$$\int \sqrt{a \operatorname{Cos}[c + d x] + b \operatorname{Sin}[c + d x]} \, dx$$

- **Derivation: Algebraic simplification**
- **Basis:** $a \operatorname{Cos}[z] + b \operatorname{Sin}[z] = \sqrt{a^2 + b^2} \operatorname{Cos}[z - \operatorname{ArcTan}[a, b]]$
- **Rule: If** $a^2 + b^2 \neq 0 \bigwedge \sqrt{a^2 + b^2} > 0$**, then**

$$\int \sqrt{a \operatorname{Cos}[c + d x] + b \operatorname{Sin}[c + d x]} \, dx \longrightarrow \left(a^2 + b^2\right)^{1/4} \int \sqrt{\operatorname{Cos}[c + d x - \operatorname{ArcTan}[a, b]]} \, dx$$

- **Program code:**

```
Int[Sqrt[a_.*Cos[c_.+d_.*x_]+b_.*Sin[c_.+d_.*x_]],x_Symbol] :=
  Dist[(a^2+b^2)^(1/4),Int[Sqrt[Cos[c+d*x-ArcTan[a,b]]],x]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2+b^2] && PositiveQ[Sqrt[a^2+b^2]]
```

- **Derivation: Piecewise constant extraction and algebraic simplification**
- **Basis:** $\partial_x \frac{\sqrt{a \operatorname{Cos}[c+d x]+b \operatorname{Sin}[c+d x]}}{\sqrt{\frac{a \operatorname{Cos}[c+d x]+b \operatorname{Sin}[c+d x]}{\sqrt{a^2+b^2}}}} = 0$
- **Basis:** $\frac{a \operatorname{Cos}[z]+b \operatorname{Sin}[z]}{\sqrt{a^2+b^2}} = \operatorname{Cos}[z - \operatorname{ArcTan}[a, b]]$
- **Rule: If** $a^2 + b^2 \neq 0 \bigwedge \neg\left(\sqrt{a^2 + b^2} > 0\right)$**, then**

$$\int \sqrt{a \operatorname{Cos}[c + d x] + b \operatorname{Sin}[c + d x]} \, dx \longrightarrow \frac{\sqrt{a \operatorname{Cos}[c + d x] + b \operatorname{Sin}[c + d x]}}{\sqrt{\frac{a \operatorname{Cos}[c+d x]+b \operatorname{Sin}[c+d x]}{\sqrt{a^2+b^2}}}} \int \sqrt{\operatorname{Cos}[c + d x - \operatorname{ArcTan}[a, b]]} \, dx$$

- **Program code:**

```
(* Int[Sqrt[a_.*Cos[c_.+d_.*x_]+b_.*Sin[c_.+d_.*x_]],x_Symbol] :=
  Sqrt[a*Cos[c+d*x]+b*Sin[c+d*x]]/Sqrt[(a*Cos[c+d*x]+b*Sin[c+d*x])/Sqrt[a^2+b^2]]*
    Int[Sqrt[Cos[c+d*x-ArcTan[a,b]]],x] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2+b^2] && Not[PositiveQ[Sqrt[a^2+b^2]]] *)
```

$$\int \frac{1}{\sqrt{a \operatorname{Cos}[c+d x]+b \operatorname{Sin}[c+d x]}} \, dx$$

- **Derivation: Algebraic simplification**
- **Basis:** $a \operatorname{Cos}[z]+b \operatorname{Sin}[z]=\sqrt{a^2+b^2} \operatorname{Cos}[z-\operatorname{ArcTan}[a, b]]$
- **Rule: If** $a^2+b^2 \neq 0 \bigwedge \sqrt{a^2+b^2}>0$, **then**

$$\int \frac{1}{\sqrt{a \operatorname{Cos}[c+d x]+b \operatorname{Sin}[c+d x]}} \, dx \longrightarrow \frac{1}{\left(a^2+b^2\right)^{1/4}} \int \frac{1}{\sqrt{\operatorname{Cos}[c+d x-\operatorname{ArcTan}[a, b]]}} \, dx$$

- **Program code:**

```
Int[1/Sqrt[a_.*Cos[c_.+d_.*x_]+b_.*Sin[c_.+d_.*x_]],x_Symbol] :=
  Dist[1/(a^2+b^2)^(1/4),Int[1/Sqrt[Cos[c+d*x-ArcTan[a,b]]],x]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2+b^2] && PositiveQ[Sqrt[a^2+b^2]]
```

- **Derivation: Piecewise constant extraction and algebraic simplification**
- **Basis:** $\partial_x \frac{\sqrt{\frac{a \operatorname{Cos}[c+d x]+b \operatorname{Sin}[c+d x]}{\sqrt{a^2+b^2}}}}{\sqrt{a \operatorname{Cos}[c+d x]+b \operatorname{Sin}[c+d x]}}=0$
- **Basis:** $\frac{a \operatorname{Cos}[z]+b \operatorname{Sin}[z]}{\sqrt{a^2+b^2}}=\operatorname{Cos}[z-\operatorname{ArcTan}[a, b]]$
- **Rule: If** $a^2+b^2 \neq 0 \bigwedge \neg\left(\sqrt{a^2+b^2}>0\right)$, **then**

$$\int \frac{1}{\sqrt{a \operatorname{Cos}[c+d x]+b \operatorname{Sin}[c+d x]}} \, dx \longrightarrow \frac{\sqrt{\frac{a \operatorname{Cos}[c+d x]+b \operatorname{Sin}[c+d x]}{\sqrt{a^2+b^2}}}}{\sqrt{a \operatorname{Cos}[c+d x]+b \operatorname{Sin}[c+d x]}} \int \frac{1}{\sqrt{\operatorname{Cos}[c+d x-\operatorname{ArcTan}[a, b]]}} \, dx$$

- **Program code:**

```
(* Int[1/Sqrt[a_.*Cos[c_.+d_.*x_]+b_.*Sin[c_.+d_.*x_]],x_Symbol] :=
  Sqrt[(a*Cos[c+d*x]+b*Sin[c+d*x])/Sqrt[a^2+b^2]]/Sqrt[a*Cos[c+d*x]+b*Sin[c+d*x]]*
    Int[1/Sqrt[Cos[c+d*x-ArcTan[a,b]]],x] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2+b^2] && Not[PositiveQ[Sqrt[a^2+b^2]]] *)
```

$\int (a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^n\,dx$

- **Rule: If $a^2+b^2=0$, then**

$$\int (a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^n\,dx \longrightarrow \frac{a\,(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^n}{b\,d\,n}$$

- **Program code:**

```
Int[(a_.*Cos[c_.+d_.*x_]+b_.*Sin[c_.+d_.*x_])^n_,x_Symbol] :=
  a*(a*Cos[c+d*x]+b*Sin[c+d*x])^n/(b*d*n) /;
FreeQ[{a,b,c,d,n},x] && ZeroQ[a^2+b^2]
```

- **Reference: G&R 2.557.5b'**
- **Rule: If $a^2+b^2\neq 0$, then**

$$\int \frac{1}{(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^2}\,dx \longrightarrow \frac{\text{Sin}[c+d\,x]}{a\,d\,(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])}$$

- **Program code:**

```
Int[1/(a_.*Cos[c_.+d_.*x_]+b_.*Sin[c_.+d_.*x_])^2,x_Symbol] :=
  Sin[c+d*x]/(a*d*(a*Cos[c+d*x]+b*Sin[c+d*x])) /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2+b^2]
```

- **Reference: G&R 2.557'**
- **Derivation: Integration by substitution**
- **Basis: If $\frac{n-1}{2}\in\mathbb{Z}$, then** $(a\,\text{Cos}[z]+b\,\text{Sin}[z])^n = \left(a^2+b^2-(-b\,\text{Cos}[z]+a\,\text{Sin}[z])^2\right)^{\frac{n-1}{2}}\,\partial_z\,(-b\,\text{Cos}[z]+a\,\text{Sin}[z])$
- **Note: Should this rule also be used for odd $n<0$?**
- **Rule: If $a^2+b^2\neq 0 \bigwedge \frac{n-1}{2}\in\mathbb{Z} \bigwedge n>0$, then**

$$\int (a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^n\,dx \longrightarrow \frac{1}{d}$$
$$\text{Subst}\left[\text{Int}\left[\left(a^2+b^2-x^2\right)^{\frac{n-1}{2}},\,x\right],\,x,\,-b\,\text{Cos}[c+d\,x]+a\,\text{Sin}[c+d\,x]\right]$$

- **Program code:**

```
Int[(a_.*Cos[c_.+d_.*x_]+b_.*Sin[c_.+d_.*x_])^n_,x_Symbol] :=
  Dist[1/d,Subst[Int[Regularize[(a^2+b^2-x^2)^((n-1)/2),x],x],x,-b*Cos[c+d*x]+a*Sin[c+d*x]]] /;
FreeQ[{a,b},x] && NonzeroQ[a^2+b^2] && OddQ[n] && n>0
```

- **Derivation: Integration by parts with a double-back flip**
- **Rule: If** $a^2+b^2\neq 0 \bigwedge n>1 \bigwedge \frac{n-1}{2}\notin\mathbb{Z}$, **then**

$$\int(a\,\mathrm{Cos}[c+d\,x]+b\,\mathrm{Sin}[c+d\,x])^n\,dx \longrightarrow$$
$$-\frac{(b\,\mathrm{Cos}[c+d\,x]-a\,\mathrm{Sin}[c+d\,x])\,(a\,\mathrm{Cos}[c+d\,x]+b\,\mathrm{Sin}[c+d\,x])^{n-1}}{d\,n}+$$
$$\frac{(n-1)\left(a^2+b^2\right)}{n}\int(a\,\mathrm{Cos}[c+d\,x]+b\,\mathrm{Sin}[c+d\,x])^{n-2}\,dx$$

- **Program code:**

```
Int[(a_.*Cos[c_.+d_.*x_]+b_.*Sin[c_.+d_.*x_])^n_,x_Symbol] :=
  -(b*Cos[c+d*x]-a*Sin[c+d*x])*(a*Cos[c+d*x]+b*Sin[c+d*x])^(n-1)/(d*n) +
  Dist[(n-1)*(a^2+b^2)/n,Int[(a*Cos[c+d*x]+b*Sin[c+d*x])^(n-2),x]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2+b^2] && RationalQ[n] && n>1 && Not[OddQ[n]]
```

- **Derivation: Integration by parts with a double-back flip**
- **Rule: If** $a^2+b^2\neq 0 \wedge n<-1 \wedge n\neq -2$, **then**

$$\int(a\,\mathrm{Cos}[c+d\,x]+b\,\mathrm{Sin}[c+d\,x])^n\,dx \longrightarrow$$
$$\frac{(b\,\mathrm{Cos}[c+d\,x]-a\,\mathrm{Sin}[c+d\,x])\,(a\,\mathrm{Cos}[c+d\,x]+b\,\mathrm{Sin}[c+d\,x])^{n+1}}{d\,(n+1)\left(a^2+b^2\right)}+$$
$$\frac{n+2}{(n+1)\left(a^2+b^2\right)}\int(a\,\mathrm{Cos}[c+d\,x]+b\,\mathrm{Sin}[c+d\,x])^{n+2}\,dx$$

- **Program code:**

```
Int[(a_.*Cos[c_.+d_.*x_]+b_.*Sin[c_.+d_.*x_])^n_,x_Symbol] :=
  (b*Cos[c+d*x]-a*Sin[c+d*x])*(a*Cos[c+d*x]+b*Sin[c+d*x])^(n+1)/(d*(n+1)*(a^2+b^2)) +
  Dist[(n+2)/((n+1)*(a^2+b^2)),Int[(a*Cos[c+d*x]+b*Sin[c+d*x])^(n+2),x]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2+b^2] && RationalQ[n] && n<-1 && n≠-2
```

$$\int \frac{\text{Cos}[c+d\,x]^m\,\text{Sin}[c+d\,x]^n}{(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^p}\,dx$$

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{\text{Cos}[z]\,\text{Sin}[z]}{a\,\text{Cos}[z]+b\,\text{Sin}[z]} = \frac{b\,\text{Cos}[z]}{a^2+b^2} + \frac{a\,\text{Sin}[z]}{a^2+b^2} - \frac{a\,b}{(a^2+b^2)\,(a\,\text{Cos}[z]+b\,\text{Sin}[z])}$
- **Rule: If** $a^2+b^2 \neq 0 \wedge m, n \in \mathbb{Z} \wedge m>0 \wedge n>0$**, then**

$$\int \frac{\text{Cos}[c+d\,x]^m\,\text{Sin}[c+d\,x]^n}{a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x]}\,dx \longrightarrow \frac{b}{a^2+b^2}\int \text{Cos}[c+d\,x]^m\,\text{Sin}[c+d\,x]^{n-1}\,dx + \frac{a}{a^2+b^2}\int \text{Cos}[c+d\,x]^{m-1}\,\text{Sin}[c+d\,x]^n\,dx - \frac{a\,b}{a^2+b^2}\int \frac{\text{Cos}[c+d\,x]^{m-1}\,\text{Sin}[c+d\,x]^{n-1}}{a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x]}\,dx$$

- **Program code:**

```
(* Int[Cos[c_.+d_.*x_]^m_.*Sin[c_.+d_.*x_]^n_./(a_.*Cos[c_.+d_.*x_]+b_.*Sin[c_.+d_.*x_]),x_Symbol] :=
  Dist[b/(a^2+b^2),Int[Cos[c+d*x]^m*Sin[c+d*x]^(n-1),x]] +
  Dist[a/(a^2+b^2),Int[Cos[c+d*x]^(m-1)*Sin[c+d*x]^n,x]] -
  Dist[a*b/(a^2+b^2),Int[Cos[c+d*x]^(m-1)*Sin[c+d*x]^(n-1)/(a*Cos[c+d*x]+b*Sin[c+d*x]),x]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2+b^2] && IntegersQ[m,n] && m>0 && n>0 *)
```

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{\text{Cos}[z]\,\text{Sin}[z]}{a\,\text{Cos}[z]+b\,\text{Sin}[z]} = \frac{b\,\text{Cos}[z]}{a^2+b^2} + \frac{a\,\text{Sin}[z]}{a^2+b^2} - \frac{a\,b}{(a^2+b^2)\,(a\,\text{Cos}[z]+b\,\text{Sin}[z])}$
- **Rule: If** $a^2+b^2 \neq 0 \wedge m, n, p \in \mathbb{Z} \wedge m>0 \wedge n>0 \wedge p<0$**, then**

$$\int \text{Cos}[c+d\,x]^m\,\text{Sin}[c+d\,x]^n\,(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^p\,dx \longrightarrow$$
$$\frac{b}{a^2+b^2}\int \text{Cos}[c+d\,x]^m\,\text{Sin}[c+d\,x]^{n-1}\,(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^{p+1}\,dx +$$
$$\frac{a}{a^2+b^2}\int \text{Cos}[c+d\,x]^{m-1}\,\text{Sin}[c+d\,x]^n\,(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^{p+1}\,dx -$$
$$\frac{a\,b}{a^2+b^2}\int \text{Cos}[c+d\,x]^{m-1}\,\text{Sin}[c+d\,x]^{n-1}\,(a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x])^p\,dx$$

- **Program code:**

```
Int[Cos[c_.+d_.*x_]^m_.*Sin[c_.+d_.*x_]^n_.*(a_.*Cos[c_.+d_.*x_]+b_.*Sin[c_.+d_.*x_])^p_,x_Symbol] :=
  Dist[b/(a^2+b^2),Int[Cos[c+d*x]^m*Sin[c+d*x]^(n-1)*(a*Cos[c+d*x]+b*Sin[c+d*x])^(p+1),x]] +
  Dist[a/(a^2+b^2),Int[Cos[c+d*x]^(m-1)*Sin[c+d*x]^n*(a*Cos[c+d*x]+b*Sin[c+d*x])^(p+1),x]] -
  Dist[a*b/(a^2+b^2),Int[Cos[c+d*x]^(m-1)*Sin[c+d*x]^(n-1)*(a*Cos[c+d*x]+b*Sin[c+d*x])^p,x]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2+b^2] && IntegersQ[m,n,p] && m>0 && n>0 && p<0
```

- **Derivation: Algebraic expansion**

- **Basis:** $\frac{\text{Sin}[z]^2}{a\,\text{Cos}[z]+b\,\text{Sin}[z]} = \frac{b\,\text{Sin}[z]}{a^2+b^2} - \frac{a\,\text{Cos}[z]}{a^2+b^2} + \frac{a^2}{(a^2+b^2)\,(a\,\text{Cos}[z]+b\,\text{Sin}[z])}$

- **Rule: If** $a^2+b^2\neq 0 \bigwedge n\in\mathbb{Z} \bigwedge n>1$**, then**

$$\int \frac{u\,\text{Sin}[c+d\,x]^n}{a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x]}\,dx \longrightarrow \frac{b}{a^2+b^2}\int u\,\text{Sin}[c+d\,x]^{n-1}\,dx - \frac{a}{a^2+b^2}\int u\,\text{Sin}[c+d\,x]^{n-2}\,\text{Cos}[c+d\,x]\,dx + \frac{a^2}{a^2+b^2}\int \frac{u\,\text{Sin}[c+d\,x]^{n-2}}{a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x]}\,dx$$

- **Program code:**

```
Int[u_.*Sin[c_.+d_.*x_]^n_./(a_.*Cos[c_.+d_.*x_]+b_.*Sin[c_.+d_.*x_]),x_Symbol] :=
  Dist[b/(a^2+b^2),Int[u*Sin[c+d*x]^(n-1),x]] -
  Dist[a/(a^2+b^2),Int[u*Sin[c+d*x]^(n-2)*Cos[c+d*x],x]] +
  Dist[a^2/(a^2+b^2),Int[u*Sin[c+d*x]^(n-2)/(a*Cos[c+d*x]+b*Sin[c+d*x]),x]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2+b^2] && IntegerQ[n] && n>0 &&
(n>1 || MatchQ[u,v_.*Tan[c+d*x]^m_. /; IntegerQ[m] && m>0])
```

- **Derivation: Algebraic expansion**

- **Basis:** $\frac{\text{Cos}[z]^2}{a\,\text{Cos}[z]+b\,\text{Sin}[z]} = \frac{a\,\text{Cos}[z]}{a^2+b^2} - \frac{b\,\text{Sin}[z]}{a^2+b^2} + \frac{b^2}{(a^2+b^2)\,(a\,\text{Cos}[z]+b\,\text{Sin}[z])}$

- **Rule: If** $a^2+b^2\neq 0 \bigwedge n\in\mathbb{Z} \bigwedge n>1$**, then**

$$\int \frac{u\,\text{Cos}[c+d\,x]^n}{a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x]}\,dx \longrightarrow \frac{a}{a^2+b^2}\int u\,\text{Cos}[c+d\,x]^{n-1}\,dx - \frac{b}{a^2+b^2}\int u\,\text{Cos}[c+d\,x]^{n-2}\,\text{Sin}[c+d\,x]\,dx + \frac{b^2}{a^2+b^2}\int \frac{u\,\text{Cos}[c+d\,x]^{n-2}}{a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x]}\,dx$$

- **Program code:**

```
Int[u_.*Cos[c_.+d_.*x_]^n_./(a_.*Cos[c_.+d_.*x_]+b_.*Sin[c_.+d_.*x_]),x_Symbol] :=
  Dist[a/(a^2+b^2),Int[u*Cos[c+d*x]^(n-1),x]] -
  Dist[b/(a^2+b^2),Int[u*Cos[c+d*x]^(n-2)*Sin[c+d*x],x]] +
  Dist[b^2/(a^2+b^2),Int[u*Cos[c+d*x]^(n-2)/(a*Cos[c+d*x]+b*Sin[c+d*x]),x]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2+b^2] && IntegerQ[n] && n>0 &&
(n>1 || MatchQ[u,v_.*Cot[c+d*x]^m_. /; IntegerQ[m] && m>0])
```

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{\text{Sec}[z]}{a\,\text{Cos}[z]+b\,\text{Sin}[z]} = \frac{\text{Tan}[z]}{b} + \frac{b\,\text{Cos}[z]-a\,\text{Sin}[z]}{b\,(a\,\text{Cos}[z]+b\,\text{Sin}[z])}$
- **Rule:**

$$\int \frac{u\,\text{Sec}[c+d\,x]}{a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x]}\,dx \longrightarrow \frac{1}{b}\int u\,\text{Tan}[c+d\,x]\,dx + \frac{1}{b}\int \frac{u\,(b\,\text{Cos}[c+d\,x]-a\,\text{Sin}[c+d\,x])}{a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x]}\,dx$$

- **Program code:**

```
(* Int[u_.*Sec[c_.+d_.*x_]/(a_.*Cos[c_.+d_.*x_]+b_.*Sin[c_.+d_.*x_]),x_Symbol] :=
  Dist[1/b,Int[u*Tan[c+d*x],x]] +
  Dist[1/b,Int[u*(b*Cos[c+d*x]-a*Sin[c+d*x])/(a*Cos[c+d*x]+b*Sin[c+d*x]),x]] /;
FreeQ[{a,b,c,d},x] *)
```

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{\text{Csc}[z]}{a\,\text{Cos}[z]+b\,\text{Sin}[z]} = \frac{\text{Cot}[z]}{a} - \frac{b\,\text{Cos}[z]-a\,\text{Sin}[z]}{a\,(a\,\text{Cos}[z]+b\,\text{Sin}[z])}$
- **Rule:**

$$\int \frac{u\,\text{Csc}[c+d\,x]}{a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x]}\,dx \longrightarrow \frac{1}{a}\int u\,\text{Cot}[c+d\,x]\,dx - \frac{1}{a}\int \frac{u\,(b\,\text{Cos}[c+d\,x]-a\,\text{Sin}[c+d\,x])}{a\,\text{Cos}[c+d\,x]+b\,\text{Sin}[c+d\,x]}\,dx$$

- **Program code:**

```
(* Int[u_.*Csc[c_.+d_.*x_]/(a_.*Cos[c_.+d_.*x_]+b_.*Sin[c_.+d_.*x_]),x_Symbol] :=
  Dist[1/a,Int[u*Cot[c+d*x],x]] -
  Dist[1/a,Int[u*(b*Cos[c+d*x]-a*Sin[c+d*x])/(a*Cos[c+d*x]+b*Sin[c+d*x]),x]] /;
FreeQ[{a,b,c,d},x] *)
```

$$\int \frac{1}{a + b\,\mathrm{Cos}[d + e\,x] + c\,\mathrm{Sin}[d + e\,x]}\,dx$$

- **Reference: G&R 2.558.4c**
- **Rule: If** $a - b = 0$**, then**

$$\int \frac{1}{a + b\,\mathrm{Cos}[d + e\,x] + c\,\mathrm{Sin}[d + e\,x]}\,dx \longrightarrow \frac{1}{c\,e}\,\mathrm{Log}\left[a + c\,\mathrm{Tan}\left[\frac{1}{2}\,(d + e\,x)\right]\right]$$

- **Program code:**

```
Int[1/(a_+b_.*Cos[d_.+e_.*x_]+c_.*Sin[d_.+e_.*x_]),x_Symbol] :=
  Log[a+c*Tan[(d+e*x)/2]]/(c*e) /;
FreeQ[{a,b,c,d,e},x] && ZeroQ[a-b]
```

- **Reference: G&R 2.558.4c**
- **Rule: If** $a + b = 0$**, then**

$$\int \frac{1}{a + b\,\mathrm{Cos}[d + e\,x] + c\,\mathrm{Sin}[d + e\,x]}\,dx \longrightarrow -\frac{1}{c\,e}\,\mathrm{Log}\left[a + c\,\mathrm{Cot}\left[\frac{1}{2}\,(d + e\,x)\right]\right]$$

- **Program code:**

```
Int[1/(a_+b_.*Cos[d_.+e_.*x_]+c_.*Sin[d_.+e_.*x_]),x_Symbol] :=
  -Log[a+c*Cot[(d+e*x)/2]]/(c*e) /;
FreeQ[{a,b,c,d,e},x] && ZeroQ[a+b]
```

- **Reference: G&R 2.558.4d**
- **Rule: If** $a^2 - b^2 - c^2 = 0$**, then**

$$\int \frac{1}{a + b\,\mathrm{Cos}[d + e\,x] + c\,\mathrm{Sin}[d + e\,x]}\,dx \longrightarrow \frac{-c + a\,\mathrm{Sin}[d + e\,x]}{c\,e\,(c\,\mathrm{Cos}[d + e\,x] - b\,\mathrm{Sin}[d + e\,x])}$$

- **Program code:**

```
Int[1/(a_+b_.*Cos[d_.+e_.*x_]+c_.*Sin[d_.+e_.*x_]),x_Symbol] :=
  (-c+a*Sin[d+e*x])/(c*e*(c*Cos[d+e*x]-b*Sin[d+e*x])) /;
FreeQ[{a,b,c,d,e},x] && ZeroQ[a^2-b^2-c^2]
```

- **Reference: G&R 2.558.4a, CRC 342b**
- **Rule: If $a^2 - b^2 \neq 0 \wedge a^2 - b^2 - c^2 > 0$, then**

$$\int \frac{1}{a + b\,\mathrm{Cos}[d + e\,x] + c\,\mathrm{Sin}[d + e\,x]}\,dx \longrightarrow \frac{2}{e\sqrt{a^2 - b^2 - c^2}}\,\mathrm{ArcTan}\left[\frac{c + (a - b)\,\mathrm{Tan}\left[\frac{1}{2}\,(d + e\,x)\right]}{\sqrt{a^2 - b^2 - c^2}}\right]$$

- **Program code:**

```
Int[1/(a_.+b_.*Cos[d_.+e_.*x_]+c_.*Sin[d_.+e_.*x_]),x_Symbol] :=
  2*ArcTan[(c+(a-b)*Tan[(d+e*x)/2])/Rt[a^2-b^2-c^2,2]]/(e*Rt[a^2-b^2-c^2,2]) /;
FreeQ[{a,b,c,d,e},x] && NonzeroQ[a^2-b^2] && PosQ[a^2-b^2-c^2]
```

- **Reference: G&R 2.558.4b', CRC 342b'**
- **Rule: If $a^2 - b^2 \neq 0 \wedge \neg\,(a^2 - b^2 - c^2 > 0)$, then**

$$\int \frac{1}{a + b\,\mathrm{Cos}[d + e\,x] + c\,\mathrm{Sin}[d + e\,x]}\,dx \longrightarrow -\frac{2}{e\sqrt{-a^2 + b^2 + c^2}}\,\mathrm{ArcTanh}\left[\frac{c + (a - b)\,\mathrm{Tan}\left[\frac{1}{2}\,(d + e\,x)\right]}{\sqrt{-a^2 + b^2 + c^2}}\right]$$

- **Program code:**

```
Int[1/(a_.+b_.*Cos[d_.+e_.*x_]+c_.*Sin[d_.+e_.*x_]),x_Symbol] :=
  -2*ArcTanh[(c+(a-b)*Tan[(d+e*x)/2])/Rt[-a^2+b^2+c^2,2]]/(e*Rt[-a^2+b^2+c^2,2]) /;
FreeQ[{a,b,c,d,e},x] && NonzeroQ[a^2-b^2] && NegQ[a^2-b^2-c^2]
```

$$\int \sqrt{a + b \operatorname{Cos}[d + e\, x] + c \operatorname{Sin}[d + e\, x]}\, dx$$

- **Reference: G&R 2.558.1 inverted with** $n = \frac{1}{2}$ **and** $a^2 - b^2 - c^2 = 0$
- **Rule: If** $a^2 - b^2 - c^2 = 0$**, then**

$$\int \sqrt{a + b \operatorname{Cos}[d + e\, x] + c \operatorname{Sin}[d + e\, x]}\, dx \longrightarrow \frac{2\, (-c \operatorname{Cos}[d + e\, x] + b \operatorname{Sin}[d + e\, x])}{e \sqrt{a + b \operatorname{Cos}[d + e\, x] + c \operatorname{Sin}[d + e\, x]}}$$

- **Program code:**

```
Int[Sqrt[a_+b_.*Cos[d_.+e_.*x_]+c_.*Sin[d_.+e_.*x_]],x_Symbol] :=
  2*(-c*Cos[d+e*x]+b*Sin[d+e*x])/(e*Sqrt[a+b*Cos[d+e*x]+c*Sin[d+e*x]]) /;
FreeQ[{a,b,c,d,e},x] && ZeroQ[a^2-b^2-c^2]
```

- **Derivation: Algebraic simplification**
- **Basis:** $a + b \operatorname{Cos}[z] + c \operatorname{Sin}[z] = a + \sqrt{b^2 + c^2} \operatorname{Cos}[z - \operatorname{ArcTan}[b, c]]$
- **Rule: If** $a^2 - b^2 - c^2 \neq 0 \bigwedge a + \sqrt{b^2 + c^2} > 0$**, then**

$$\int \sqrt{a + b \operatorname{Cos}[d + e\, x] + c \operatorname{Sin}[d + e\, x]}\, dx \longrightarrow \int \sqrt{a + \sqrt{b^2 + c^2} \operatorname{Cos}[d + e\, x - \operatorname{ArcTan}[b, c]]}\, dx$$

- **Program code:**

```
Int[Sqrt[a_+b_.*Cos[d_.+e_.*x_]+c_.*Sin[d_.+e_.*x_]],x_Symbol] :=
  Int[Sqrt[a+Sqrt[b^2+c^2]*Cos[d+e*x-ArcTan[b,c]]],x] /;
FreeQ[{a,b,c,d,e},x] && NonzeroQ[a^2-b^2-c^2] && PositiveQ[a+Sqrt[b^2+c^2]]
```

- **Derivation: Piecewise constant extraction and algebraic simplification**

- **Basis:** $\partial_x \frac{\sqrt{a+b\,\text{Cos}[d+e\,x]+c\,\text{Sin}[d+e\,x]}}{\sqrt{\frac{a+b\,\text{Cos}[d+e\,x]+c\,\text{Sin}[d+e\,x]}{a+\sqrt{b^2+c^2}}}} = 0$

- **Basis:** $a + b\,\text{Cos}[z] + c\,\text{Sin}[z] = a + \sqrt{b^2+c^2}\,\text{Cos}[z - \text{ArcTan}[b, c]]$

- **Rule:** If $a^2 - b^2 - c^2 \neq 0 \bigwedge \neg \left(a + \sqrt{b^2+c^2} > 0\right)$, then

$$\int \sqrt{a+b\,\text{Cos}[d+e\,x]+c\,\text{Sin}[d+e\,x]}\,dx \longrightarrow \frac{\sqrt{a+b\,\text{Cos}[d+e\,x]+c\,\text{Sin}[d+e\,x]}}{\sqrt{\frac{a+b\,\text{Cos}[d+e\,x]+c\,\text{Sin}[d+e\,x]}{a+\sqrt{b^2+c^2}}}}$$

$$\int \sqrt{\frac{a}{a+\sqrt{b^2+c^2}} + \frac{\sqrt{b^2+c^2}}{a+\sqrt{b^2+c^2}}\,\text{Cos}[d+e\,x-\text{ArcTan}[b, c]]}\,dx$$

- **Program code:**

```
Int[Sqrt[a_.+b_.*Cos[d_.+e_.*x_]+c_.*Sin[d_.+e_.*x_]],x_Symbol] :=
  Sqrt[a+b*Cos[d+e*x]+c*Sin[d+e*x]]/Sqrt[(a+b*Cos[d+e*x]+c*Sin[d+e*x])/(a+Sqrt[b^2+c^2])]*
    Int[Sqrt[a/(a+Sqrt[b^2+c^2])+Sqrt[b^2+c^2]/(a+Sqrt[b^2+c^2])*Cos[d+e*x-ArcTan[b,c]]],x] /;
FreeQ[{a,b,c,d,e},x] && NonzeroQ[a^2-b^2-c^2] && Not[PositiveQ[a+Sqrt[b^2+c^2]]]
```

$$\int \frac{1}{\sqrt{a + b\,\mathrm{Cos}[d + e\,x] + c\,\mathrm{Sin}[d + e\,x]}}\,dx$$

- **Derivation: Algebraic simplification   NonzeroQ[a^2 - b^2 - c^2] ????**

- **Basis:** $a + b\,\mathrm{Cos}[z] + c\,\mathrm{Sin}[z] = a + \sqrt{b^2 + c^2}\,\mathrm{Cos}[z - \mathrm{ArcTan}[b, c]]$

- **Rule: If** $a + \sqrt{b^2 + c^2} > 0$**, then**

$$\int \frac{1}{\sqrt{a + b\,\mathrm{Cos}[d + e\,x] + c\,\mathrm{Sin}[d + e\,x]}}\,dx \longrightarrow \int \frac{1}{\sqrt{a + \sqrt{b^2 + c^2}\,\mathrm{Cos}[d + e\,x - \mathrm{ArcTan}[b, c]]}}\,dx$$

- **Program code:**

```
Int[1/Sqrt[a_+b_.*Cos[d_.+e_.*x_]+c_.*Sin[d_.+e_.*x_]],x_Symbol] :=
  Int[1/Sqrt[a+Sqrt[b^2+c^2]*Cos[d+e*x-ArcTan[b,c]]],x] /;
FreeQ[{a,b,c,d,e},x] && PositiveQ[a+Sqrt[b^2+c^2]]
```

- **Derivation: Piecewise constant extraction and algebraic simplification**

- **Basis:** $\partial_x \frac{\sqrt{\frac{a+b\,\mathrm{Cos}[d+e\,x]+c\,\mathrm{Sin}[d+e\,x]}{a+\sqrt{b^2+c^2}}}}{\sqrt{a+b\,\mathrm{Cos}[d+e\,x]+c\,\mathrm{Sin}[d+e\,x]}} = 0$

- **Basis:** $a + b\,\mathrm{Cos}[z] + c\,\mathrm{Sin}[z] = a + \sqrt{b^2 + c^2}\,\mathrm{Cos}[z - \mathrm{ArcTan}[b, c]]$

- **Rule: If** $a + \sqrt{b^2 + c^2} \neq 0 \bigwedge \neg \left(a + \sqrt{b^2 + c^2} > 0\right)$**, then**

$$\int \frac{1}{\sqrt{a + b\,\mathrm{Cos}[d + e\,x] + c\,\mathrm{Sin}[d + e\,x]}}\,dx \longrightarrow \frac{\sqrt{\frac{a+b\,\mathrm{Cos}[d+e\,x]+c\,\mathrm{Sin}[d+e\,x]}{a+\sqrt{b^2+c^2}}}}{\sqrt{a + b\,\mathrm{Cos}[d + e\,x] + c\,\mathrm{Sin}[d + e\,x]}}$$
$$\int \frac{1}{\sqrt{\frac{a}{a+\sqrt{b^2+c^2}} + \frac{\sqrt{b^2+c^2}}{a+\sqrt{b^2+c^2}}\,\mathrm{Cos}[d + e\,x - \mathrm{ArcTan}[b, c]]}}\,dx$$

- **Program code:**

```
Int[1/Sqrt[a_.+b_.*Cos[d_.+e_.*x_]+c_.*Sin[d_.+e_.*x_]],x_Symbol] :=
  Sqrt[(a+b*Cos[d+e*x]+c*Sin[d+e*x])/(a+Sqrt[b^2+c^2])]/Sqrt[a+b*Cos[d+e*x]+c*Sin[d+e*x]]*
    Int[1/Sqrt[a/(a+Sqrt[b^2+c^2])+Sqrt[b^2+c^2]/(a+Sqrt[b^2+c^2])*Cos[d+e*x-ArcTan[b,c]]],x] /;
FreeQ[{a,b,c,d,e},x] && NonzeroQ[a+Sqrt[b^2+c^2]] && Not[PositiveQ[a+Sqrt[b^2+c^2]]]
```

$$\int (a + b\,\mathrm{Cos}[d + e\,x] + c\,\mathrm{Sin}[d + e\,x])^n\, dx$$

- **Reference: G&R 2.558.1 inverted with** $a^2 - b^2 - c^2 = 0$
- **Rule: If** $a^2 - b^2 - c^2 = 0 \bigwedge n \in \mathbb{F} \bigwedge n > 1$**, then**

$$\int (a + b\,\mathrm{Cos}[d + e\,x] + c\,\mathrm{Sin}[d + e\,x])^n\, dx \longrightarrow$$
$$\frac{(-c\,\mathrm{Cos}[d + e\,x] + b\,\mathrm{Sin}[d + e\,x])\,(a + b\,\mathrm{Cos}[d + e\,x] + c\,\mathrm{Sin}[d + e\,x])^{n-1}}{e\,n} +$$
$$\frac{a\,(2\,n - 1)}{n} \int (a + b\,\mathrm{Cos}[d + e\,x] + c\,\mathrm{Sin}[d + e\,x])^{n-1}\, dx$$

- **Program code:**

```
Int[(a_+b_.*Cos[d_.+e_.*x_]+c_.*Sin[d_.+e_.*x_])^n_,x_Symbol] :=
  (-c*Cos[d+e*x]+b*Sin[d+e*x])*(a+b*Cos[d+e*x]+c*Sin[d+e*x])^(n-1)/(e*n) +
  Dist[a*(2*n-1)/n,Int[(a+b*Cos[d+e*x]+c*Sin[d+e*x])^(n-1),x]] /;
FreeQ[{a,b,c,d,e},x] && ZeroQ[a^2-b^2-c^2] && RationalQ[n] && n>1
```

- **Reference: G&R 2.558.1 inverted**
- **Rule: If** $a^2 - b^2 - c^2 \neq 0 \bigwedge n \in \mathbb{F} \bigwedge n > 1$**, then**

$$\int (a + b\,\mathrm{Cos}[d + e\,x] + c\,\mathrm{Sin}[d + e\,x])^n\, dx \longrightarrow$$
$$\frac{(-c\,\mathrm{Cos}[d + e\,x] + b\,\mathrm{Sin}[d + e\,x])\,(a + b\,\mathrm{Cos}[d + e\,x] + c\,\mathrm{Sin}[d + e\,x])^{n-1}}{e\,n} +$$
$$\frac{1}{n} \int \left(n\,a^2 + (n - 1)\,\left(b^2 + c^2\right) + a\,b\,(2\,n - 1)\,\mathrm{Cos}[d + e\,x] + a\,c\,(2\,n - 1)\,\mathrm{Sin}[d + e\,x]\right)$$
$$(a + b\,\mathrm{Cos}[d + e\,x] + c\,\mathrm{Sin}[d + e\,x])^{n-2}\, dx$$

- **Program code:**

```
Int[(a_+b_.*Cos[d_.+e_.*x_]+c_.*Sin[d_.+e_.*x_])^n_,x_Symbol] :=
  (-c*Cos[d+e*x]+b*Sin[d+e*x])*(a+b*Cos[d+e*x]+c*Sin[d+e*x])^(n-1)/(e*n) +
  Dist[1/n,Int[(n*a^2+(n-1)*(b^2+c^2)+a*b*(2*n-1)*Cos[d+e*x]+a*c*(2*n-1)*Sin[d+e*x])*
    (a+b*Cos[d+e*x]+c*Sin[d+e*x])^(n-2),x]] /;
FreeQ[{a,b,c,d,e},x] && NonzeroQ[a^2-b^2-c^2] && FractionQ[n] && n>1
```

$$\int \frac{1}{(a + b\,\text{Cos}[d + e\,x] + c\,\text{Sin}[d + e\,x])^n}\,dx$$

- **Reference: G&R 2.558.1 inverted with $a^2 - b^2 - c^2 = 0$ inverted**
- **Rule: If $a^2 - b^2 - c^2 = 0 \land n < -1$, then**

$$\int (a + b\,\text{Cos}[d + e\,x] + c\,\text{Sin}[d + e\,x])^n\,dx \longrightarrow$$
$$\frac{(c\,\text{Cos}[d + e\,x] - b\,\text{Sin}[d + e\,x])\,(a + b\,\text{Cos}[d + e\,x] + c\,\text{Sin}[d + e\,x])^n}{a\,e\,(2\,n + 1)} +$$
$$\frac{n + 1}{a\,(2\,n + 1)} \int (a + b\,\text{Cos}[d + e\,x] + c\,\text{Sin}[d + e\,x])^{n+1}\,dx$$

- **Program code:**

```
Int[(a_+b_.*Cos[d_.+e_.*x_]+c_.*Sin[d_.+e_.*x_])^n_,x_Symbol] :=
  (c*Cos[d+e*x]-b*Sin[d+e*x])*(a+b*Cos[d+e*x]+c*Sin[d+e*x])^n/(a*e*(2*n+1)) +
  Dist[(n+1)/(a*(2*n+1)),Int[(a+b*Cos[d+e*x]+c*Sin[d+e*x])^(n+1),x]] /;
FreeQ[{a,b,c,d,e},x] && ZeroQ[a^2-b^2-c^2] && RationalQ[n] && n<-1
```

- **Reference: G&R 2.558.1 with $n = -2$**
- **Rule: If $a^2 - b^2 - c^2 \neq 0$, then**

$$\int \frac{1}{(a + b\,\text{Cos}[d + e\,x] + c\,\text{Sin}[d + e\,x])^2}\,dx \longrightarrow$$
$$\frac{c\,\text{Cos}[d + e\,x] - b\,\text{Sin}[d + e\,x]}{e\,\left(a^2 - b^2 - c^2\right)\,(a + b\,\text{Cos}[d + e\,x] + c\,\text{Sin}[d + e\,x])} +$$
$$\frac{a}{a^2 - b^2 - c^2} \int \frac{1}{a + b\,\text{Cos}[d + e\,x] + c\,\text{Sin}[d + e\,x]}\,dx$$

- **Program code:**

```
Int[1/(a_+b_.*Cos[d_.+e_.*x_]+c_.*Sin[d_.+e_.*x_])^2,x_Symbol] :=
  (c*Cos[d+e*x]-b*Sin[d+e*x])/(e*(a^2-b^2-c^2)*(a+b*Cos[d+e*x]+c*Sin[d+e*x])) +
  Dist[a/(a^2-b^2-c^2),Int[1/(a+b*Cos[d+e*x]+c*Sin[d+e*x]),x]] /;
FreeQ[{a,b,c,d,e},x] && NonzeroQ[a^2-b^2-c^2]
```

- **Reference: G&R 2.558.1 with** $n = -\frac{3}{2}$

- **Rule: If** $a^2 - b^2 - c^2 \neq 0$**, then**

$$\int \frac{1}{(a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x])^{3/2}} \, dx \longrightarrow \frac{2\,(c \operatorname{Cos}[d + e x] - b \operatorname{Sin}[d + e x])}{e\,\left(a^2 - b^2 - c^2\right) \sqrt{a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x]}} + \frac{1}{a^2 - b^2 - c^2} \int \sqrt{a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x]} \, dx$$

- **Program code:**

```
Int[1/(a_+b_.*Cos[d_.+e_.*x_]+c_.*Sin[d_.+e_.*x_])^(3/2),x_Symbol] :=
  2*(c*Cos[d+e*x]-b*Sin[d+e*x])/(e*(a^2-b^2-c^2)*Sqrt[a+b*Cos[d+e*x]+c*Sin[d+e*x]]) +
  Dist[1/(a^2-b^2-c^2),Int[Sqrt[a+b*Cos[d+e*x]+c*Sin[d+e*x]],x]] /;
FreeQ[{a,b,c,d,e},x] && NonzeroQ[a^2-b^2-c^2]
```

- **Reference: G&R 2.558.1**

- **Rule: If** $a^2 - b^2 - c^2 \neq 0 \bigwedge n < -1 \bigwedge n \neq -2 \bigwedge n \neq -\frac{3}{2}$**, then**

$$\int (a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x])^n \, dx \longrightarrow \frac{(-c \operatorname{Cos}[d + e x] + b \operatorname{Sin}[d + e x])\,(a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x])^{n+1}}{e\,(n + 1)\,\left(a^2 - b^2 - c^2\right)} + \frac{1}{(n + 1)\,\left(a^2 - b^2 - c^2\right)} \int ((n + 1)\,a - (n + 2)\,b \operatorname{Cos}[d + e x] - (n + 2)\,c \operatorname{Sin}[d + e x])\,(a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x])^{n+1} \, dx$$

- **Program code:**

```
Int[(a_+b_.*Cos[d_.+e_.*x_]+c_.*Sin[d_.+e_.*x_])^n_,x_Symbol] :=
  (-c*Cos[d+e*x]+b*Sin[d+e*x])*(a+b*Cos[d+e*x]+c*Sin[d+e*x])^(n+1)/(e*(n+1)*(a^2-b^2-c^2)) +
  Dist[1/((n+1)*(a^2-b^2-c^2)),
    Int[((n+1)*a-(n+2)*b*Cos[d+e*x]-(n+2)*c*Sin[d+e*x])*(a+b*Cos[d+e*x]+c*Sin[d+e*x])^(n+1),x]] /;
FreeQ[{a,b,c,d,e},x] && NonzeroQ[a^2-b^2-c^2] && RationalQ[n] && n<-1 && n≠-2 && n≠-3/2
```

$\int (A + B\,Cos[d + e\,x] + C\,Sin[d + e\,x])\,(a + b\,Cos[d + e\,x] + c\,Sin[d + e\,x])^n\,dx$

- **Note: Although exactly analogous to G&R 2.451.3 for hyperbolic functions, there is no corresponding G&R 2.558.n formula for trig functions. Apparently the authors did not anticipate $b^2 + c^2$ could be 0 in the complex plane.**

- **Rule: If $b^2 + c^2 = 0$, then**

$$\int \frac{A + B\,Cos[d + e\,x] + C\,Sin[d + e\,x]}{a + b\,Cos[d + e\,x] + c\,Sin[d + e\,x]}\,dx \longrightarrow \frac{(2\,a\,A - b\,B - c\,C)\,x}{2\,a^2} - \frac{(b\,B + c\,C)\,(b\,Cos[d + e\,x] - c\,Sin[d + e\,x])}{2\,a\,b\,c\,e} + \frac{\left(a^2\,(b\,B - c\,C) - 2\,a\,A\,b^2 + b^2\,(b\,B + c\,C)\right)\,Log[a + b\,Cos[d + e\,x] + c\,Sin[d + e\,x]]}{2\,a^2\,b\,c\,e}$$

- **Program code:**

```
Int[(A_.+B_.*Cos[d_.+e_.*x_]+C_.*Sin[d_.+e_.*x_])/(a_+b_.*Cos[d_.+e_.*x_]+c_.*Sin[d_.+e_.*x_]),x_Sym
  (2*a*A-b*B-c*C)*x/(2*a^2) - (b*B+c*C)*(b*Cos[d+e*x]-c*Sin[d+e*x])/(2*a*b*c*e) +
  (a^2*(b*B-c*C)-2*a*A*b^2+b^2*(b*B+c*C))*Log[a+b*Cos[d+e*x]+c*Sin[d+e*x]]/(2*a^2*b*c*e) /;
FreeQ[{a,b,c,d,e,A,B,C},x] && ZeroQ[b^2+c^2]
```

```
Int[(A_.+C_.*Sin[d_.+e_.*x_])/(a_+b_.*Cos[d_.+e_.*x_]+c_.*Sin[d_.+e_.*x_]),x_Symbol] :=
  (2*a*A-c*C)*x/(2*a^2) - C*Cos[d+e*x]/(2*a*e) + c*C*Sin[d+e*x]/(2*a*b*e) +
  (-a^2*C+2*a*c*A+b^2*C)*Log[a+b*Cos[d+e*x]+c*Sin[d+e*x]]/(2*a^2*b*e) /;
FreeQ[{a,b,c,d,e,A,C},x] && ZeroQ[b^2+c^2]
```

```
Int[(A_.+B_.*Cos[d_.+e_.*x_])/(a_+b_.*Cos[d_.+e_.*x_]+c_.*Sin[d_.+e_.*x_]),x_Symbol] :=
  (2*a*A-b*B)*x/(2*a^2) - b*B*Cos[d+e*x]/(2*a*c*e) + B*Sin[d+e*x]/(2*a*e) +
  (a^2*B-2*a*b*A+b^2*B)*Log[a+b*Cos[d+e*x]+c*Sin[d+e*x]]/(2*a^2*c*e) /;
FreeQ[{a,b,c,d,e,A,B},x] && ZeroQ[b^2+c^2]
```

- **Reference: G&R 2.558.2 with $A\,(b^2 + c^2) - a\,(b\,B + c\,C) = 0$**

- **Rule: If $b^2 + c^2 \neq 0 \wedge A\,(b^2 + c^2) - a\,(b\,B + c\,C) = 0$, then**

$$\int \frac{A + B\,Cos[d + e\,x] + C\,Sin[d + e\,x]}{a + b\,Cos[d + e\,x] + c\,Sin[d + e\,x]}\,dx \longrightarrow \frac{(b\,B + c\,C)\,x}{b^2 + c^2} + \frac{(c\,B - b\,C)\,Log[a + b\,Cos[d + e\,x] + c\,Sin[d + e\,x]]}{e\,\left(b^2 + c^2\right)}$$

- **Program code:**

```
Int[(A_.+B_.*Cos[d_.+e_.*x_]+C_.*Sin[d_.+e_.*x_])/(a_.+b_.*Cos[d_.+e_.*x_]+c_.*Sin[d_.+e_.*x_]),x_Sy
  (b*B+c*C)*x/(b^2+c^2) + (c*B-b*C)*Log[a+b*Cos[d+e*x]+c*Sin[d+e*x]]/(e*(b^2+c^2)) /;
FreeQ[{a,b,c,d,e,A,B,C},x] && NonzeroQ[b^2+c^2] && ZeroQ[A*(b^2+c^2)-a*(b*B+c*C)]
```

- **Reference: G&R 2.558.2 with** $B = 0$ **and** $A\,(b^2 + c^2) - a\,c\,C = 0$

```
Int[(A_.+C_.*Sin[d_.+e_.*x_])/(a_.+b_.*Cos[d_.+e_.*x_]+c_.*Sin[d_.+e_.*x_]),x_Symbol] :=
  c*C*x/(b^2+c^2) - b*C*Log[a+b*Cos[d+e*x]+c*Sin[d+e*x]]/(e*(b^2+c^2)) /;
FreeQ[{a,b,c,d,e,A,C},x] && NonzeroQ[b^2+c^2] && ZeroQ[A*(b^2+c^2)-a*c*C]
```

- **Reference: G&R 2.558.2 with** $C = 0$ **and** $A\,(b^2 + c^2) - a\,b\,B = 0$

```
Int[(A_.+B_.*Cos[d_.+e_.*x_])/(a_.+b_.*Cos[d_.+e_.*x_]+c_.*Sin[d_.+e_.*x_]),x_Symbol] :=
  b*B*x/(b^2+c^2) + c*B*Log[a+b*Cos[d+e*x]+c*Sin[d+e*x]]/(e*(b^2+c^2)) /;
FreeQ[{a,b,c,d,e,A,B},x] && NonzeroQ[b^2+c^2] && ZeroQ[A*(b^2+c^2)-a*b*B]
```

- **Reference: G&R 2.558.2**
- **Rule: If** $b^2 + c^2 \neq 0 \land A\,(b^2 + c^2) - a\,(b\,B + c\,C) \neq 0$**, then**

$$\int \frac{A + B\,\mathrm{Cos}[d + e\,x] + C\,\mathrm{Sin}[d + e\,x]}{a + b\,\mathrm{Cos}[d + e\,x] + c\,\mathrm{Sin}[d + e\,x]}\,dx \longrightarrow \frac{(b\,B + c\,C)\,x}{b^2 + c^2} + \frac{(c\,B - b\,C)\,\mathrm{Log}[a + b\,\mathrm{Cos}[d + e\,x] + c\,\mathrm{Sin}[d + e\,x]]}{e\,(b^2 + c^2)} + \frac{A\,(b^2 + c^2) - a\,(b\,B + c\,C)}{b^2 + c^2}\int \frac{1}{a + b\,\mathrm{Cos}[d + e\,x] + c\,\mathrm{Sin}[d + e\,x]}\,dx$$

- **Program code:**

```
Int[(A_.+B_.*Cos[d_.+e_.*x_]+C_.*Sin[d_.+e_.*x_])/(a_.+b_.*Cos[d_.+e_.*x_]+c_.*Sin[d_.+e_.*x_]),x_Sy
  (b*B+c*C)*x/(b^2+c^2) + (c*B-b*C)*Log[a+b*Cos[d+e*x]+c*Sin[d+e*x]]/(e*(b^2+c^2)) +
  Dist[(A*(b^2+c^2)-a*(b*B+c*C))/(b^2+c^2),Int[1/(a+b*Cos[d+e*x]+c*Sin[d+e*x]),x]] /;
FreeQ[{a,b,c,d,e,A,B,C},x] && NonzeroQ[b^2+c^2] && NonzeroQ[A*(b^2+c^2)-a*(b*B+c*C)]
```

- **Reference: G&R 2.558.2 with** $B = 0$

```
Int[(A_.+C_.*Sin[d_.+e_.*x_])/(a_.+b_.*Cos[d_.+e_.*x_]+c_.*Sin[d_.+e_.*x_]),x_Symbol] :=
  c*C*(d+e*x)/(e*(b^2+c^2)) - b*C*Log[a+b*Cos[d+e*x]+c*Sin[d+e*x]]/(e*(b^2+c^2)) +
  Dist[(A*(b^2+c^2)-a*c*C)/(b^2+c^2),Int[1/(a+b*Cos[d+e*x]+c*Sin[d+e*x]),x]] /;
FreeQ[{a,b,c,d,e,A,C},x] && NonzeroQ[b^2+c^2] && NonzeroQ[A*(b^2+c^2)-a*c*C]
```

- **Reference: G&R 2.558.2 with** $C = 0$

```
Int[(A_.+B_.*Cos[d_.+e_.*x_])/(a_.+b_.*Cos[d_.+e_.*x_]+c_.*Sin[d_.+e_.*x_]),x_Symbol] :=
  b*B*(d+e*x)/(e*(b^2+c^2)) +
  c*B*Log[a+b*Cos[d+e*x]+c*Sin[d+e*x]]/(e*(b^2+c^2)) +
  Dist[(A*(b^2+c^2)-a*b*B)/(b^2+c^2),Int[1/(a+b*Cos[d+e*x]+c*Sin[d+e*x]),x]] /;
FreeQ[{a,b,c,d,e,A,B},x] && NonzeroQ[b^2+c^2] && NonzeroQ[A*(b^2+c^2)-a*b*B]
```

- **Reference: G&R 2.558.1 with $n = -2$ and $a A - b B - c C = 0$**
- **Rule: If $a^2 - b^2 - c^2 \neq 0 \land a A - b B - c C = 0$, then**

$$\int \frac{A + B \operatorname{Cos}[d + e x] + C \operatorname{Sin}[d + e x]}{(a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x])^2} \, dx \longrightarrow \frac{c B - b C - (a C - c A) \operatorname{Cos}[d + e x] + (a B - b A) \operatorname{Sin}[d + e x]}{e \left(a^2 - b^2 - c^2\right) (a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x])}$$

- **Program code:**

```
Int[(A_.+B_.*Cos[d_.+e_.*x_]+C_.*Sin[d_.+e_.*x_])/(a_.+b_.*Cos[d_.+e_.*x_]+c_.*Sin[d_.+e_.*x_])^2,x_
  (c*B-b*C-(a*C-c*A)*Cos[d+e*x]+(a*B-b*A)*Sin[d+e*x])/
    (e*(a^2-b^2-c^2)*(a+b*Cos[d+e*x]+c*Sin[d+e*x])) /;
FreeQ[{a,b,c,d,e,A,B,C},x] && NonzeroQ[a^2-b^2-c^2] && ZeroQ[a*A-b*B-c*C]
```

- **Reference: G&R 2.558.1 with $B = 0$, $n = -2$ and $a A - c C = 0$**

```
Int[(A_.+C_.*Sin[d_.+e_.*x_])/(a_.+b_.*Cos[d_.+e_.*x_]+c_.*Sin[d_.+e_.*x_])^2,x_Symbol] :=
  -(b*C+(a*C-c*A)*Cos[d+e*x]+b*A*Sin[d+e*x])/(e*(a^2-b^2-c^2)*(a+b*Cos[d+e*x]+c*Sin[d+e*x])) /;
FreeQ[{a,b,c,d,e,A,C},x] && NonzeroQ[a^2-b^2-c^2] && ZeroQ[a*A-c*C]
```

- **Reference: G&R 2.558.1 with $C = 0$, $n = -2$ and $a A - b B = 0$**

```
Int[(A_.+B_.*Cos[d_.+e_.*x_])/(a_.+b_.*Cos[d_.+e_.*x_]+c_.*Sin[d_.+e_.*x_])^2,x_Symbol] :=
  (c*B+c*A*Cos[d+e*x]+(a*B-b*A)*Sin[d+e*x])/(e*(a^2-b^2-c^2)*(a+b*Cos[d+e*x]+c*Sin[d+e*x])) /;
FreeQ[{a,b,c,d,e,A,B},x] && NonzeroQ[a^2-b^2-c^2] && ZeroQ[a*A-b*B]
```

- **Reference: G&R 2.558.1 with $n = -2$**
- **Rule: If $a^2 - b^2 - c^2 \neq 0 \land a A - b B - c C \neq 0$, then**

$$\int \frac{A + B \operatorname{Cos}[d + e x] + C \operatorname{Sin}[d + e x]}{(a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x])^2} \, dx \longrightarrow$$
$$\frac{c B - b C - (a C - c A) \operatorname{Cos}[d + e x] + (a B - b A) \operatorname{Sin}[d + e x]}{e \left(a^2 - b^2 - c^2\right) (a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x])} +$$
$$\frac{a A - b B - c C}{a^2 - b^2 - c^2} \int \frac{1}{a + b \operatorname{Cos}[d + e x] + c \operatorname{Sin}[d + e x]} \, dx$$

- **Program code:**

```
Int[(A_.+B_.*Cos[d_.+e_.*x_]+C_.*Sin[d_.+e_.*x_])/(a_.+b_.*Cos[d_.+e_.*x_]+c_.*Sin[d_.+e_.*x_])^2,x_
  (c*B-b*C-(a*C-c*A)*Cos[d+e*x]+(a*B-b*A)*Sin[d+e*x])/
    (e*(a^2-b^2-c^2)*(a+b*Cos[d+e*x]+c*Sin[d+e*x])) +
  Dist[(a*A-b*B-c*C)/(a^2-b^2-c^2),Int[1/(a+b*Cos[d+e*x]+c*Sin[d+e*x]),x]] /;
FreeQ[{a,b,c,d,e,A,B,C},x] && NonzeroQ[a^2-b^2-c^2] && NonzeroQ[a*A-b*B-c*C]
```

- **Reference: G&R 2.558.1 with** $B = 0$ **and** $n = -2$

```
Int[(A_.+C_.*Sin[d_.+e_.*x_])/(a_.+b_.*Cos[d_.+e_.*x_]+c_.*Sin[d_.+e_.*x_])^2,x_Symbol] :=
  -(b*C+(a*C-c*A)*Cos[d+e*x]+b*A*Sin[d+e*x])/(e*(a^2-b^2-c^2)*(a+b*Cos[d+e*x]+c*Sin[d+e*x])) +
  Dist[(a*A-c*C)/(a^2-b^2-c^2),Int[1/(a+b*Cos[d+e*x]+c*Sin[d+e*x]),x]] /;
FreeQ[{a,b,c,d,e,A,C},x] && NonzeroQ[a^2-b^2-c^2] && NonzeroQ[a*A-c*C]
```

- **Reference: G&R 2.558.1 with** $C = 0$ **and** $n = -2$

```
Int[(A_.+B_.*Cos[d_.+e_.*x_])/(a_.+b_.*Cos[d_.+e_.*x_]+c_.*Sin[d_.+e_.*x_])^2,x_Symbol] :=
  (c*B+c*A*Cos[d+e*x]+(a*B-b*A)*Sin[d+e*x])/(e*(a^2-b^2-c^2)*(a+b*Cos[d+e*x]+c*Sin[d+e*x])) +
  Dist[(a*A-b*B)/(a^2-b^2-c^2),Int[1/(a+b*Cos[d+e*x]+c*Sin[d+e*x]),x]] /;
FreeQ[{a,b,c,d,e,A,B},x] && NonzeroQ[a^2-b^2-c^2] && NonzeroQ[a*A-b*B]
```

- **Reference: G&R 2.558.1**
- **Rule: If** $a^2 - b^2 - c^2 \neq 0 \land n < -1 \land n \neq -2$**, then**

$$\int (A + B\,\text{Cos}[d + e\,x] + C\,\text{Sin}[d + e\,x])\,(a + b\,\text{Cos}[d + e\,x] + c\,\text{Sin}[d + e\,x])^n\,dx \longrightarrow$$

$$-\frac{(c\,B - b\,C - (a\,C - c\,A)\,\text{Cos}[d + e\,x] + (a\,B - b\,A)\,\text{Sin}[d + e\,x])\,(a + b\,\text{Cos}[d + e\,x] + c\,\text{Sin}[d + e\,x])^{n+1}}{e\,(n + 1)\,\left(a^2 - b^2 - c^2\right)} +$$

$$\frac{1}{(n + 1)\,\left(a^2 - b^2 - c^2\right)}$$

$$\int ((n + 1)\,(a\,A - b\,B - c\,C) + (n + 2)\,(a\,B - b\,A)\,\text{Cos}[d + e\,x] + (n + 2)\,(a\,C - c\,A)\,\text{Sin}[d + e\,x])$$
$$(a + b\,\text{Cos}[d + e\,x] + c\,\text{Sin}[d + e\,x])^{n+1}\,dx$$

- **Program code:**

```
Int[(A_.+B_.*Cos[d_.+e_.*x_]+C_.*Sin[d_.+e_.*x_])*(a_.+b_.*Cos[d_.+e_.*x_]+c_.*Sin[d_.+e_.*x_])^n_,x
  -(c*B-b*C-(a*C-c*A)*Cos[d+e*x]+(a*B-b*A)*Sin[d+e*x])*(a+b*Cos[d+e*x]+c*Sin[d+e*x])^(n+1)/
    (e*(n+1)*(a^2-b^2-c^2)) +
  Dist[1/((n+1)*(a^2-b^2-c^2)),
    Int[((n+1)*(a*A-b*B-c*C)+(n+2)*(a*B-b*A)*Cos[d+e*x]+(n+2)*(a*C-c*A)*Sin[d+e*x])*
      (a+b*Cos[d+e*x]+c*Sin[d+e*x])^(n+1),x]] /;
FreeQ[{a,b,c,d,e,A,B,C},x] && NonzeroQ[a^2-b^2-c^2] && RationalQ[n] && n<-1 && n≠-2
```

- **Reference: G&R 2.558.1 with** $B = 0$

```
Int[(A_.+C_.*Sin[d_.+e_.*x_])*(a_.+b_.*Cos[d_.+e_.*x_]+c_.*Sin[d_.+e_.*x_])^n_,x_Symbol] :=
  (b*C+(a*C-c*A)*Cos[d+e*x]+b*A*Sin[d+e*x])*(a+b*Cos[d+e*x]+c*Sin[d+e*x])^(n+1)/
    (e*(n+1)*(a^2-b^2-c^2)) +
  Dist[1/((n+1)*(a^2-b^2-c^2)),
    Int[((n+1)*(a*A-c*C)-(n+2)*b*A*Cos[d+e*x]+(n+2)*(a*C-c*A)*Sin[d+e*x])*
      (a+b*Cos[d+e*x]+c*Sin[d+e*x])^(n+1),x]] /;
FreeQ[{a,b,c,d,e,A,C},x] && NonzeroQ[a^2-b^2-c^2] && RationalQ[n] && n<-1 && n≠-2
```

- **Reference: G&R 2.558.1 with** $C = 0$

```
Int[(A_.+B_.*Cos[d_.+e_.*x_])*(a_.+b_.*Cos[d_.+e_.*x_]+c_.*Sin[d_.+e_.*x_])^n_,x_Symbol] :=
  -(c*B+c*A*Cos[d+e*x]+(a*B-b*A)*Sin[d+e*x])*(a+b*Cos[d+e*x]+c*Sin[d+e*x])^(n+1)/
    (e*(n+1)*(a^2-b^2-c^2)) +
  Dist[1/((n+1)*(a^2-b^2-c^2)),
    Int[((n+1)*(a*A-b*B)+(n+2)*(a*B-b*A)*Cos[d+e*x]-(n+2)*c*A*Sin[d+e*x])*
      (a+b*Cos[d+e*x]+c*Sin[d+e*x])^(n+1),x]] /;
FreeQ[{a,b,c,d,e,A,B},x] && NonzeroQ[a^2-b^2-c^2] && RationalQ[n] && n<-1 && n≠-2
```

- **Derivation: Algebraic simplification**
- **Basis:** $(A+B\,z)\,(a+b\,z)^n = \frac{B}{b}\,(a+b\,z)^{n+1} + \frac{(A\,b-a\,B)}{b}\,(a+b\,z)^n$
- **Rule: If** $b\,C - c\,B = 0 \bigwedge b\,A - a\,B \neq 0 \bigwedge \left(n = -\frac{1}{2} \bigvee a^2 - b^2 - c^2 = 0\right)$**, then**

$$\int (A + B\,\mathrm{Cos}[d+e\,x] + C\,\mathrm{Sin}[d+e\,x])\,(a+b\,\mathrm{Cos}[d+e\,x]+c\,\mathrm{Sin}[d+e\,x])^n\,dx \longrightarrow$$
$$\frac{B}{b}\int (a+b\,\mathrm{Cos}[d+e\,x]+c\,\mathrm{Sin}[d+e\,x])^{n+1}\,dx + \frac{b\,A-a\,B}{b}\int (a+b\,\mathrm{Cos}[d+e\,x]+c\,\mathrm{Sin}[d+e\,x])^n\,dx$$

- **Program code:**

```
Int[(A_.+B_.*Cos[d_.+e_.*x_]+C_.*Sin[d_.+e_.*x_])*(a_+b_.*Cos[d_.+e_.*x_]+c_.*Sin[d_.+e_.*x_])^n_,x_
  Dist[B/b,Int[(a+b*Cos[d+e*x]+c*Sin[d+e*x])^(n+1),x]] +
  Dist[(b*A-a*B)/b,Int[(a+b*Cos[d+e*x]+c*Sin[d+e*x])^n,x]] /;
FreeQ[{a,b,c,d,e,A,B,C},x] && ZeroQ[b*C-c*B] && NonzeroQ[b*A-a*B] && RationalQ[n] && (n==-1/2 || Zer
```

- **Reference: G&R 2.558.1 inverted**
- **Rule: If** $a^2 - b^2 - c^2 \neq 0 \bigwedge n \in \mathbb{F} \bigwedge n > 0$**, then**

$$\int (A + B\,\mathrm{Cos}[d+e\,x] + C\,\mathrm{Sin}[d+e\,x])\,(a+b\,\mathrm{Cos}[d+e\,x]+c\,\mathrm{Sin}[d+e\,x])^n\,dx \longrightarrow$$
$$\frac{(B\,c - b\,C - a\,C\,\mathrm{Cos}[d+e\,x] + a\,B\,\mathrm{Sin}[d+e\,x])\,(a+b\,\mathrm{Cos}[d+e\,x]+c\,\mathrm{Sin}[d+e\,x])^n}{a\,e\,(n+1)} +$$
$$\frac{1}{a\,(n+1)}\int \Big(a\,(b\,B+c\,C)\,n + a^2\,A\,(n+1) + \left(a^2\,B\,n + c\,(b\,C-c\,B)\,n + a\,b\,A\,(n+1)\right)\mathrm{Cos}[d+e\,x] +$$
$$\left(a^2\,C\,n - b\,(b\,C-c\,B)\,n + a\,c\,A\,(n+1)\right)\mathrm{Sin}[d+e\,x]\Big)\,(a+b\,\mathrm{Cos}[d+e\,x]+c\,\mathrm{Sin}[d+e\,x])^{n-1}\,dx$$

- **Program code:**

```
Int[(A_.+B_.*Cos[d_.+e_.*x_]+C_.*Sin[d_.+e_.*x_])*(a_+b_.*Cos[d_.+e_.*x_]+c_.*Sin[d_.+e_.*x_])^n_,x_
  (B*c-b*C-a*C*Cos[d+e*x]+a*B*Sin[d+e*x])*(a+b*Cos[d+e*x]+c*Sin[d+e*x])^n/(a*e*(n+1)) +
  Dist[1/(a*(n+1)),
    Int[(a*(b*B+c*C)*n + a^2*A*(n+1) +
        (a^2*B*n + c*(b*C-c*B)*n + a*b*A*(n+1))*Cos[d+e*x] +
        (a^2*C*n - b*(b*C-c*B)*n + a*c*A*(n+1))*Sin[d+e*x])*
      (a+b*Cos[d+e*x]+c*Sin[d+e*x])^(n-1),x]] /;
FreeQ[{a,b,c,d,e,A,B,C},x] && NonzeroQ[a^2-b^2-c^2] && RationalQ[n] && n>0
```

Reference: G&R 2.558.1 inverted with B = 0

```
Int[(A_.+C_.*Sin[d_.+e_.*x_])*(a_+b_.*Cos[d_.+e_.*x_]+c_.*Sin[d_.+e_.*x_])^n_,x_Symbol] :=
  -(b*C+a*C*Cos[d+e*x])*(a+b*Cos[d+e*x]+c*Sin[d+e*x])^n/(a*e*(n+1)) +
  Dist[1/(a*(n+1)),
    Int[(a*c*C*n+a^2*A*(n+1)+(c*b*C*n+a*b*A*(n+1))*Cos[d+e*x]+(a^2*C*n-b^2*C*n+a*c*A*(n+1))*Sin[d+e*
      (a+b*Cos[d+e*x]+c*Sin[d+e*x])^(n-1),x]] /;
FreeQ[{a,b,c,d,e,A,C},x] && NonzeroQ[a^2-b^2-c^2] && RationalQ[n] && n>0
```

- **Reference: G&R 2.558.1 inverted with C = 0**

```
Int[(A_.+B_.*Cos[d_.+e_.*x_])*(a_+b_.*Cos[d_.+e_.*x_]+c_.*Sin[d_.+e_.*x_])^n_,x_Symbol] :=
  (B*c+a*B*Sin[d+e*x])*(a+b*Cos[d+e*x]+c*Sin[d+e*x])^n/(a*e*(n+1)) +
  Dist[1/(a*(n+1)),
    Int[(a*b*B*n+a^2*A*(n+1)+(a^2*B*n-c^2*B*n+a*b*A*(n+1))*Cos[d+e*x]+(b*c*B*n+a*c*A*(n+1))*Sin[d+e*
      (a+b*Cos[d+e*x]+c*Sin[d+e*x])^(n-1),x]] /;
FreeQ[{a,b,c,d,e,A,B},x] && NonzeroQ[a^2-b^2-c^2] && RationalQ[n] && n>0
```