$$\int \frac{1}{\sqrt{a + b\,x + c\,x^2}}\,dx$$

- **Reference: G&R 2.261.3 which is correct only for $b + 2\,c\,x > 0$**
- **Derivation: Piecewise constant extraction**
- **Basis: If $b^2 - 4\,a\,c = 0$, then $\partial_x \frac{b+2\,c\,x}{\sqrt{a+b\,x+c\,x^2}} = 0$**
- **Rule: If $b^2 - 4\,a\,c = 0$, then**

$$\int \frac{1}{\sqrt{a + b\,x + c\,x^2}}\,dx \longrightarrow \frac{b + 2\,c\,x}{\sqrt{a + b\,x + c\,x^2}} \int \frac{1}{b + 2\,c\,x}\,dx$$

- **Program code:**

```
Int[1/Sqrt[a_+b_.*x_+c_.*x_^2],x_Symbol] :=
  (b+2*c*x)/Sqrt[a+b*x+c*x^2]*Int[1/(b+2*c*x),x] /;
FreeQ[{a,b,c},x] && ZeroQ[b^2-4*a*c]
```

- **Reference: G&R 2.261.2, CRC 237b, A&S 3.3.34**
- **Derivation: Primitive rule**
- **Basis: $\partial_x \text{ArcSinh}[x] = \frac{1}{\sqrt{1+x^2}}$**
- **Note: Unlike the formulas in the references, this rule is valid even if c is not positive.**
- **Rule: If $4\,a - \frac{b^2}{c} > 0 \bigwedge c > 0$, then**

$$\int \frac{1}{\sqrt{a + b\,x + c\,x^2}}\,dx \longrightarrow \frac{1}{\sqrt{c}}\,\text{ArcSinh}\left[\frac{b + 2\,c\,x}{\sqrt{c}\,\sqrt{4\,a - \frac{b^2}{c}}}\right]$$

- **Program code:**

```
Int[1/Sqrt[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  ArcSinh[(b+2*c*x)/(Rt[c,2]*Sqrt[4*a-b^2/c])]/Rt[c,2] /;
FreeQ[{a,b,c},x] && PositiveQ[4*a-b^2/c] && PosQ[c]
```

- **Reference: G&R 2.261.3, CRC 238, A&S 3.3.36**
- **Derivation: Primitive rule**
- **Basis:** $\partial_x \text{ArcSin}[x] = \frac{1}{\sqrt{1-x^2}}$
- **Note: Unlike the formulas in the references, this rule is valid even if c is not positive.**
- **Rule: If** $4a - \frac{b^2}{c} > 0 \bigwedge \neg (c > 0)$**, then**

$$\int \frac{1}{\sqrt{a+b x+c x^2}} \, dx \longrightarrow -\frac{1}{\sqrt{-c}} \text{ArcSin}\left[\frac{b+2 c x}{\sqrt{-c} \sqrt{4 a-\frac{b^2}{c}}}\right]$$

- **Program code:**

```
Int[1/Sqrt[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  -ArcSin[(b+2*c*x)/(Rt[-c,2]*Sqrt[4*a-b^2/c])]/Rt[-c,2] /;
FreeQ[{a,b,c},x] && PositiveQ[4*a-b^2/c] && NegQ[c]
```

- **Rule: If** $\neg \left(4a - \frac{b^2}{c} > 0\right) \bigwedge c > 0$**, then**

$$\int \frac{1}{\sqrt{b x+c x^2}} \, dx \longrightarrow \frac{2}{\sqrt{c}} \text{ArcTanh}\left[\frac{\sqrt{c}\, x}{\sqrt{b x+c x^2}}\right]$$

- **Program code:**

```
Int[1/Sqrt[b_.*x_+c_.*x_^2],x_Symbol] :=
  2*ArcTanh[Rt[c,2]*x/Sqrt[b*x+c*x^2]]/Rt[c,2] /;
FreeQ[{b,c},x] && Not[PositiveQ[-b^2/c]] && PosQ[c]
```

- **Rule: If** $\neg \left(4a - \frac{b^2}{c} > 0\right) \bigwedge \neg (c > 0)$**, then**

$$\int \frac{1}{\sqrt{b x+c x^2}} \, dx \longrightarrow \frac{2}{\sqrt{-c}} \text{ArcTan}\left[\frac{\sqrt{-c}\, x}{\sqrt{b x+c x^2}}\right]$$

- **Program code:**

```
Int[1/Sqrt[b_.*x_+c_.*x_^2],x_Symbol] :=
  2*ArcTan[Rt[-c,2]*x/Sqrt[b*x+c*x^2]]/Rt[-c,2] /;
FreeQ[{b,c},x] && Not[PositiveQ[-b^2/c]] && NegQ[c]
```

- **Reference: G&R 2.261.1, CRC 237a, A&S 3.3.33**
- **Derivation: Primitive rule**
- **Rule: If $b^2-4\,a\,c\neq 0\,\wedge\,c>0$, then**

$$\int\frac{1}{\sqrt{a+b\,x+c\,x^2}}\,dx\,\longrightarrow\,\frac{1}{\sqrt{c}}\,\text{ArcTanh}\Big[\frac{b+2\,c\,x}{2\,\sqrt{c}\,\sqrt{a+b\,x+c\,x^2}}\Big]$$

- **Program code:**

```
Int[1/Sqrt[a_+b_.*x_+c_.*x_^2],x_Symbol] :=
  ArcTanh[(b+2*c*x)/(2*Rt[c,2]*Sqrt[a+b*x+c*x^2])]/Rt[c,2] /;
FreeQ[{a,b,c},x] && NonzeroQ[b^2-4*a*c] && PosQ[c]
```

- **Reference: CRC 238**
- **Derivation: Primitive rule**
- **Rule: If $b^2-4\,a\,c\neq 0\,\wedge\,\neg\,(c>0)$, then**

$$\int\frac{1}{\sqrt{a+b\,x+c\,x^2}}\,dx\,\longrightarrow\,-\frac{1}{\sqrt{-c}}\,\text{ArcTan}\Big[\frac{b+2\,c\,x}{2\,\sqrt{-c}\,\sqrt{a+b\,x+c\,x^2}}\Big]$$

- **Program code:**

```
Int[1/Sqrt[a_+b_.*x_+c_.*x_^2],x_Symbol] :=
  -ArcTan[(b+2*c*x)/(2*Rt[-c,2]*Sqrt[a+b*x+c*x^2])]/Rt[-c,2] /;
FreeQ[{a,b,c},x] && NonzeroQ[b^2-4*a*c] && NegQ[c]
```

$$\int \left(a + b\,x + c\,x^2\right)^n \, dx$$

- **Rule: If** $b^2 - 4\,a\,c = 0 \land 2\,n + 1 \neq 0 \land n \notin \mathbb{Z}$, **then**

$$\int \left(a + b\,x + c\,x^2\right)^n dx \longrightarrow \frac{(b + 2\,c\,x)\,\left(a + b\,x + c\,x^2\right)^n}{2\,c\,(2\,n + 1)}$$

- **Program code:**

```
Int[(a_+b_.*x_+c_.*x_^2)^n_,x_Symbol] :=
  (b+2*c*x)*(a+b*x+c*x^2)^n/(2*c*(2*n+1)) /;
FreeQ[{a,b,c,n},x] && ZeroQ[b^2-4*a*c] && NonzeroQ[2*n+1] && Not[IntegerQ[n]]
```

- **Reference: G&R 2.264.5, CRC 239**
- **Rule: If** $b^2 - 4\,a\,c \neq 0$, **then**

$$\int \frac{1}{\left(a + b\,x + c\,x^2\right)^{3/2}} dx \longrightarrow -\frac{2\,(b + 2\,c\,x)}{\left(b^2 - 4\,a\,c\right)\sqrt{a + b\,x + c\,x^2}}$$

- **Program code:**

```
Int[1/(a_.+b_.*x_+c_.*x_^2)^(3/2),x_Symbol] :=
  -2*(b+2*c*x)/((b^2-4*a*c)*Sqrt[a+b*x+c*x^2]) /;
FreeQ[{a,b,c},x] && NonzeroQ[b^2-4*a*c]
```

- **Reference: G&R 2.260.2, CRC 245, A&S 3.3.37**
- **Rule: If** $b^2 - 4\,a\,c \neq 0 \land n \in \mathbb{F} \land n > 0$, **then**

$$\int \left(a + b\,x + c\,x^2\right)^n dx \longrightarrow \frac{(b + 2\,c\,x)\,\left(a + b\,x + c\,x^2\right)^n}{2\,c\,(2\,n + 1)} - \frac{n\,\left(b^2 - 4\,a\,c\right)}{2\,c\,(2\,n + 1)} \int \left(a + b\,x + c\,x^2\right)^{n-1} dx$$

- **Program code:**

```
Int[(a_.+b_.*x_+c_.*x_^2)^n_,x_Symbol] :=
  (b+2*c*x)*(a+b*x+c*x^2)^n/(2*c*(2*n+1)) -
  Dist[n*(b^2-4*a*c)/(2*c*(2*n+1)),Int[(a+b*x+c*x^2)^(n-1),x]] /;
FreeQ[{a,b,c},x] && NonzeroQ[b^2-4*a*c] && FractionQ[n] && n>0
```

- **Reference: G&R 2.263.3, CRC 241**

- **Rule: If** $b^2-4\,a\,c\neq 0\,\wedge\,n\in\mathbb{F}\,\wedge\,n<-1$**, then**

$$\int\left(a+b\,x+c\,x^2\right)^n\,dx\;\longrightarrow\;\frac{(b+2\,c\,x)\,\left(a+b\,x+c\,x^2\right)^{n+1}}{(n+1)\,\left(b^2-4\,a\,c\right)}-\frac{2\,c\,(2\,n+3)}{(n+1)\,\left(b^2-4\,a\,c\right)}\int\left(a+b\,x+c\,x^2\right)^{n+1}\,dx$$

- **Program code:**

```
Int[(a_.+b_.*x_+c_.*x_^2)^n_,x_Symbol] :=
  (b+2*c*x)*(a+b*x+c*x^2)^(n+1)/((n+1)*(b^2-4*a*c)) -
  Dist[2*c*(2*n+3)/((n+1)*(b^2-4*a*c)),Int[(a+b*x+c*x^2)^(n+1),x]] /;
FreeQ[{a,b,c},x] && NonzeroQ[b^2-4*a*c] && FractionQ[n] && n<-1
```

$$\int \frac{1}{(d+e\,x)\,\sqrt{a+c\,x^2}}\,dx$$

- **Reference: G&R 2.266.7, CRC 260**
- **Note: This is an unnecessary special case of the integration rule for $(d+e\,x)^m\,(a+c\,x^2)^n$ when $m+2\,(n+1)=0$.**
- **Rule: If $c\,d^2+a\,e^2=0$, then**

$$\int \frac{1}{(d+e\,x)\,\sqrt{a+c\,x^2}}\,dx \longrightarrow \frac{e\,\sqrt{a+c\,x^2}}{c\,d\,(d+e\,x)}$$

- **Program code:**

```
(* Int[1/((d_+e_.*x_)*Sqrt[a_.+c_.*x_^2]),x_Symbol] :=
   e*Sqrt[a+c*x^2]/(c*d*(d+e*x)) /;
FreeQ[{a,c,d,e},x] && ZeroQ[c*d^2+a*e^2] *)
```

- **Reference: G&R 2.266.1, CRC 258**
- **Rule: If $c\,d^2+a\,e^2>0$, then**

$$\int \frac{1}{(d+e\,x)\,\sqrt{a+c\,x^2}}\,dx \longrightarrow -\frac{1}{\sqrt{c\,d^2+a\,e^2}}\,\mathrm{ArcTanh}\Big[\frac{a\,e-c\,d\,x}{\sqrt{c\,d^2+a\,e^2}\,\sqrt{a+c\,x^2}}\Big]$$

- **Program code:**

```
Int[1/((d_+e_.*x_)*Sqrt[a_.+c_.*x_^2]),x_Symbol] :=
  -ArcTanh[(a*e-c*d*x)/(Rt[c*d^2+a*e^2,2]*Sqrt[a+c*x^2])]/Rt[c*d^2+a*e^2,2] /;
FreeQ[{a,c,d,e},x] && PosQ[c*d^2+a*e^2]
```

- **Reference: G&R 2.266.3, CRC 259**
- **Rule: If $\neg\,(c\,d^2+a\,e^2>0)$, then**

$$\int \frac{1}{(d+e\,x)\,\sqrt{a+c\,x^2}}\,dx \longrightarrow \frac{1}{\sqrt{-c\,d^2-a\,e^2}}\,\mathrm{ArcTan}\Big[\frac{a\,e-c\,d\,x}{\sqrt{-c\,d^2-a\,e^2}\,\sqrt{a+c\,x^2}}\Big]$$

- **Program code:**

```
Int[1/((d_+e_.*x_)*Sqrt[a_.+c_.*x_^2]),x_Symbol] :=
  ArcTan[(a*e-c*d*x)/(Rt[-c*d^2-a*e^2,2]*Sqrt[a+c*x^2])]/Rt[-c*d^2-a*e^2,2] /;
FreeQ[{a,c,d,e},x] && NegQ[c*d^2+a*e^2]
```

$$\int \frac{1}{(d+e\,x)\,\sqrt{a+b\,x+c\,x^2}}\,dx$$

- **Reference: G&R 2.266.7, CRC 260**
- **Rule: If** $c\,d^2-b\,d\,e+a\,e^2=0\,\wedge\,2\,c\,d-b\,e\neq 0$**, then**

$$\int \frac{1}{(d+e\,x)\,\sqrt{a+b\,x+c\,x^2}}\,dx\;\longrightarrow\;\frac{2\,e\,\sqrt{a+b\,x+c\,x^2}}{(2\,c\,d-b\,e)\,(d+e\,x)}$$

- **Program code:**

```
Int[1/((d_.+e_.*x_)*Sqrt[a_.+b_.*x_+c_.*x_^2]),x_Symbol] :=
  2*e*Sqrt[a+b*x+c*x^2]/((2*c*d-b*e)*(d+e*x)) /;
FreeQ[{a,b,c,d,e},x] && ZeroQ[c*d^2-b*d*e+a*e^2] && NonzeroQ[2*c*d-b*e]
```

- **Reference: G&R 2.266.6 which is correct only for** $2\,a+b\,x>0$
- **Derivation: Piecewise constant extraction**
- **Basis: If** $b^2-4\,a\,c=0$**, then** $\partial_x \frac{b+2\,c\,x}{\sqrt{a+b\,x+c\,x^2}}=0$
- **Rule: If** $b^2-4\,a\,c=0$**, then**

$$\int \frac{1}{(d+e\,x)\,\sqrt{a+b\,x+c\,x^2}}\,dx\;\longrightarrow\;\frac{b+2\,c\,x}{\sqrt{a+b\,x+c\,x^2}}\int \frac{1}{(d+e\,x)\,(b+2\,c\,x)}\,dx$$

- **Program code:**

```
Int[1/((d_.+e_.*x_)*Sqrt[a_+b_.*x_+c_.*x_^2]),x_Symbol] :=
  (b+2*c*x)/Sqrt[a+b*x+c*x^2]*Int[1/((d+e*x)*(b+2*c*x)),x] /;
FreeQ[{a,b,c,d,e},x] && ZeroQ[b^2-4*a*c]
```

- **Rule: If** $2\,c\,d-b\,e=0\bigwedge\frac{b^2-4\,a\,c}{c}>0$**, then**

$$\int\frac{1}{(d+e\,x)\,\sqrt{a+b\,x+c\,x^2}}\,dx\;\longrightarrow\;\frac{2}{e\,\sqrt{\frac{b^2-4\,a\,c}{c}}}\,\text{ArcTan}\Big[\frac{2\,\sqrt{a+b\,x+c\,x^2}}{\sqrt{\frac{b^2-4\,a\,c}{c}}}\Big]$$

- **Program code:**

```
Int[1/((d_.+e_.*x_)*Sqrt[a_.+b_.*x_+c_.*x_^2]),x_Symbol] :=
  2/(e*Rt[(b^2-4*a*c)/c,2])*ArcTan[2*Sqrt[a+b*x+c*x^2]/Rt[(b^2-4*a*c)/c,2]] /;
FreeQ[{a,b,c,d,e},x] && ZeroQ[2*c*d-b*e] && PosQ[(b^2-4*a*c)/c]
```

- **Rule: If** $2\,c\,d-b\,e=0\bigwedge\neg\left(\frac{b^2-4\,a\,c}{c}>0\right)$**, then**

$$\int\frac{1}{(d+e\,x)\,\sqrt{a+b\,x+c\,x^2}}\,dx\;\longrightarrow\;-\frac{2}{e\,\sqrt{\frac{4\,a\,c-b^2}{c}}}\,\text{ArcTanh}\Big[\frac{2\,\sqrt{a+b\,x+c\,x^2}}{\sqrt{\frac{4\,a\,c-b^2}{c}}}\Big]$$

- **Program code:**

```
Int[1/((d_.+e_.*x_)*Sqrt[a_.+b_.*x_+c_.*x_^2]),x_Symbol] :=
  -2/(e*Rt[(4*a*c-b^2)/c,2])*ArcTanh[2*Sqrt[a+b*x+c*x^2]/Rt[(4*a*c-b^2)/c,2]] /;
FreeQ[{a,b,c,d,e},x] && ZeroQ[2*c*d-b*e] && NegQ[(b^2-4*a*c)/c]
```

- **Reference: G&R 2.266.1, CRC 258**
- **Rule: If** $b^2-4\,a\,c\neq0\,\wedge\,2\,c\,d-b\,e\neq0\,\wedge\,c\,d^2-b\,d\,e+a\,e^2>0$**, then**

$$\int\frac{1}{(d+e\,x)\,\sqrt{a+b\,x+c\,x^2}}\,dx\;\longrightarrow\;-\frac{1}{\sqrt{c\,d^2-b\,d\,e+a\,e^2}}\,\text{ArcTanh}\Big[\frac{2\,a\,e-b\,d-(2\,c\,d-b\,e)\,x}{2\,\sqrt{c\,d^2-b\,d\,e+a\,e^2}\,\sqrt{a+b\,x+c\,x^2}}\Big]$$

- **Program code:**

```
Int[1/((d_.+e_.*x_)*Sqrt[a_.+b_.*x_+c_.*x_^2]),x_Symbol] :=
  -1/Rt[c*d^2-b*d*e+a*e^2,2]*
  ArcTanh[(2*a*e-b*d-(2*c*d-b*e)*x)/(2*Rt[c*d^2-b*d*e+a*e^2,2]*Sqrt[a+b*x+c*x^2])] /;
FreeQ[{a,b,c,d,e},x] && NonzeroQ[b^2-4*a*c] && NonzeroQ[2*c*d-b*e] && PosQ[c*d^2-b*d*e+a*e^2]
```

- **Reference: G&R 2.266.3, CRC 259**

- **Rule: If** $b^2 - 4 a c \neq 0 \wedge 2 c d - b e \neq 0 \wedge \neg (c d^2 - b d e + a e^2 > 0)$**, then**

$$\int \frac{1}{(d + e x) \sqrt{a + b x + c x^2}} \, dx \longrightarrow \frac{1}{\sqrt{-c d^2 + b d e - a e^2}} \operatorname{ArcTan}\left[\frac{2 a e - b d - (2 c d - b e) x}{2 \sqrt{-c d^2 + b d e - a e^2} \sqrt{a + b x + c x^2}}\right]$$

- **Program code:**

```
Int[1/((d_.+e_.*x_)*Sqrt[a_.+b_.*x_+c_.*x_^2]),x_Symbol] :=
  1/Rt[-c*d^2+b*d*e-a*e^2,2]*
  ArcTan[(2*a*e-b*d-(2*c*d-b*e)*x)/(2*Rt[-c*d^2+b*d*e-a*e^2,2]*Sqrt[a+b*x+c*x^2])] /;
FreeQ[{a,b,c,d,e},x] && NonzeroQ[b^2-4*a*c] && NonzeroQ[2*c*d-b*e] && NegQ[c*d^2-b*d*e+a*e^2]
```

$$\int \frac{\left(a+b\,x+c\,x^2\right)^n}{d+e\,x}\,dx$$

- **Reference: G&R 2.265b**

- **Rule: If** $n \in \mathbb{F} \wedge n>0 \wedge c\,d^2-b\,d\,e+a\,e^2=0$, **then**

$$\int \frac{\left(a+b\,x+c\,x^2\right)^n}{d+e\,x}\,dx \longrightarrow \frac{\left(a+b\,x+c\,x^2\right)^n}{2\,e\,n}-\frac{2\,c\,d-b\,e}{2\,e^2}\int \left(a+b\,x+c\,x^2\right)^{n-1}dx$$

- **Program code:**

```
Int[(a_.+b_.*x_+c_.*x_^2)^n_/(d_.+e_.*x_),x_Symbol] :=
  (a+b*x+c*x^2)^n/(2*e*n) -
  Dist[(2*c*d-b*e)/(2*e^2),Int[(a+b*x+c*x^2)^(n-1),x]] /;
FreeQ[{a,b,c,d,e},x] && FractionQ[n] && n>0 && ZeroQ[c*d^2-b*d*e+a*e^2]
```

- **Reference: G&R 2.265b**

- **Rule: If** $n \in \mathbb{F} \wedge n>0 \wedge 2\,c\,d-b\,e=0$, **then**

$$\int \frac{\left(a+b\,x+c\,x^2\right)^n}{d+e\,x}\,dx \longrightarrow \frac{\left(a+b\,x+c\,x^2\right)^n}{2\,e\,n}+\frac{c\,d^2-b\,d\,e+a\,e^2}{e^2}\int \frac{\left(a+b\,x+c\,x^2\right)^{n-1}}{d+e\,x}\,dx$$

- **Program code:**

```
Int[(a_.+b_.*x_+c_.*x_^2)^n_/(d_.+e_.*x_),x_Symbol] :=
  (a+b*x+c*x^2)^n/(2*e*n) +
  Dist[(c*d^2-b*d*e+a*e^2)/e^2,Int[(a+b*x+c*x^2)^(n-1)/(d+e*x),x]] /;
FreeQ[{a,b,c,d,e},x] && FractionQ[n] && n>0 && ZeroQ[2*c*d-b*e]
```

- **Reference: G&R 2.265b**

- **Rule: If** $n \in \mathbb{F} \wedge n > 0$**, then**

$$\int \frac{\left(a+b\,x+c\,x^2\right)^n}{d+e\,x}\,dx \;\longrightarrow\; \frac{\left(a+b\,x+c\,x^2\right)^n}{2\,e\,n} - \frac{2\,c\,d-b\,e}{2\,e^2}\int \left(a+b\,x+c\,x^2\right)^{n-1}dx + \frac{c\,d^2-b\,d\,e+a\,e^2}{e^2}\int \frac{\left(a+b\,x+c\,x^2\right)^{n-1}}{d+e\,x}\,dx$$

- **Program code:**

```
Int[(a_.+b_.*x_+c_.*x_^2)^n_/(d_.+e_.*x_),x_Symbol] :=
  (a+b*x+c*x^2)^n/(2*e*n) -
  Dist[(2*c*d-b*e)/(2*e^2), Int[(a+b*x+c*x^2)^(n-1),x]] +
  Dist[(c*d^2-b*d*e+a*e^2)/e^2,Int[(a+b*x+c*x^2)^(n-1)/(d+e*x),x]] /;
FreeQ[{a,b,c,d,e},x] && FractionQ[n] && n>0
```

- **Reference: G&R 2.268b, CRC 122**

- **Rule: If** $n \in \mathbb{F} \wedge n < -1 \wedge c\,d^2-b\,d\,e+a\,e^2 \neq 0 \wedge b\,e-2\,c\,d = 0$**, then**

$$\int \frac{\left(a+b\,x+c\,x^2\right)^n}{d+e\,x}\,dx \;\longrightarrow\; -\frac{e\left(a+b\,x+c\,x^2\right)^{n+1}}{2\,(n+1)\left(c\,d^2-b\,d\,e+a\,e^2\right)} + \frac{e^2}{c\,d^2-b\,d\,e+a\,e^2}\int \frac{\left(a+b\,x+c\,x^2\right)^{n+1}}{d+e\,x}\,dx$$

- **Program code:**

```
Int[(a_.+b_.*x_+c_.*x_^2)^n_/(d_.+e_.*x_),x_Symbol] :=
  -e*(a+b*x+c*x^2)^(n+1)/(2*(n+1)*(c*d^2-b*d*e+a*e^2)) +
  Dist[e^2/(c*d^2-b*d*e+a*e^2),Int[(a+b*x+c*x^2)^(n+1)/(d+e*x),x]] /;
FreeQ[{a,b,c,d,e},x] && FractionQ[n] && n<-1 && NonzeroQ[c*d^2-b*d*e+a*e^2] && ZeroQ[2*c*d-b*e]
```

- **Reference: G&R 2.268b, CRC 122**

- **Rule: If** $n \in \mathbb{F} \wedge n < -1 \wedge c d^2 - b d e + a e^2 \neq 0$**, then**

$$\int \frac{\left(a + b x + c x^2\right)^n}{d + e x} \, dx \longrightarrow -\frac{e \left(a + b x + c x^2\right)^{n+1}}{2 (n+1) \left(c d^2 - b d e + a e^2\right)} + \frac{2 c d - b e}{2 \left(c d^2 - b d e + a e^2\right)} \int \left(a + b x + c x^2\right)^n dx + \frac{e^2}{c d^2 - b d e + a e^2} \int \frac{\left(a + b x + c x^2\right)^{n+1}}{d + e x} \, dx$$

- **Program code:**

```
Int[(a_.+b_.*x_+c_.*x_^2)^n_/(d_.+e_.*x_),x_Symbol] :=
  -e*(a+b*x+c*x^2)^(n+1)/(2*(n+1)*(c*d^2-b*d*e+a*e^2)) +
  Dist[(2*c*d-b*e)/(2*(c*d^2-b*d*e+a*e^2)), Int[(a+b*x+c*x^2)^n,x]] +
  Dist[e^2/(c*d^2-b*d*e+a*e^2),Int[(a+b*x+c*x^2)^(n+1)/(d+e*x),x]] /;
FreeQ[{a,b,c,d,e},x] && FractionQ[n] && n<-1 && NonzeroQ[c*d^2-b*d*e+a*e^2]
```

$$\int \frac{1}{\sqrt{a+b\,x^2+c\,x^4}}\,dx$$

- **Derivation: Algebraic expansion and piecewise constant extraction**
- **Basis: If** $q=\sqrt{b^2-4\,a\,c}$ **, then** $a+b\,x^2+c\,x^4=a\left(1+\frac{2\,c\,x^2}{b-q}\right)\left(1+\frac{2\,c\,x^2}{b+q}\right)$
- **Basis: If** $q=\sqrt{b^2-4\,a\,c}$ **, then** $\partial_x\frac{\sqrt{1+\frac{2\,c\,x^2}{b-q}}\,\sqrt{1+\frac{2\,c\,x^2}{b+q}}}{\sqrt{a+b\,x^2+c\,x^4}}=0$
- **Rule: If** $b^2-4\,a\,c\neq 0$**, let** $q=\sqrt{b^2-4\,a\,c}$ **, then**

$$\int \frac{1}{\sqrt{a+b\,x^2+c\,x^4}}\,dx\to\frac{\sqrt{1+\frac{2\,c\,x^2}{b-q}}\,\sqrt{1+\frac{2\,c\,x^2}{b+q}}}{\sqrt{a+b\,x^2+c\,x^4}}\int\frac{1}{\sqrt{1+\frac{2\,c\,x^2}{b-q}}\,\sqrt{1+\frac{2\,c\,x^2}{b+q}}}\,dx$$

- **Program code:**

```
Int[1/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
  Module[{q=Rt[b^2-4*a*c,2]},
  Sqrt[1+2*c*x^2/(b-q)]*Sqrt[1+2*c*x^2/(b+q)]/Sqrt[a+b*x^2+c*x^4]*
  Int[1/(Sqrt[1+2*c*x^2/(b-q)]*Sqrt[1+2*c*x^2/(b+q)]),x]] /;
FreeQ[{a,b,c},x] && NonzeroQ[b^2-4*a*c]
```

$$\int \frac{d + e\,x^2}{\sqrt{a + b\,x^2 + c\,x^4}}\,dx$$

- **Derivation: Piecewise constant extraction**
- **Basis: If** $b^2 - 4\,a\,c = 0$, **then** $\partial_x \frac{b+2\,c\,x^2}{\sqrt{a+b\,x^2+c\,x^4}} = 0$
- **Rule: If** $b^2 - 4\,a\,c = 0$, **then**

$$\int \frac{d + e\,x^2}{\sqrt{a + b\,x^2 + c\,x^4}}\,dx \longrightarrow \frac{b + 2\,c\,x^2}{\sqrt{a + b\,x^2 + c\,x^4}} \int \frac{d + e\,x^2}{b + 2\,c\,x^2}\,dx$$

- **Program code:**

```
Int[(d_.+e_.*x_^2)/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
  Dist[(b+2*c*x^2)/Sqrt[a+b*x^2+c*x^4],Int[(d+e*x^2)/(b+2*c*x^2),x]] /;
FreeQ[{a,b,c,d,e},x] && ZeroQ[b^2-4*a*c]
```

- **Derivation: Algebraic expansion**
- **Basis: If** $a > 0$, **let** $q = \sqrt{b^2 - 4\,a\,c}$, **then** $\sqrt{a + b\,x^2 + c\,x^4} = \sqrt{a}\,\sqrt{1 + \frac{2\,c\,x^2}{b-q}}\,\sqrt{1 + \frac{2\,c\,x^2}{b+q}}$
- **Rule: If** $b^2 - 4\,a\,c \neq 0 \bigwedge a > 0$, **let** $q = \sqrt{b^2 - 4\,a\,c}$, **then**

$$\int \frac{d + e\,x^2}{\sqrt{a + b\,x^2 + c\,x^4}}\,dx \longrightarrow \frac{1}{\sqrt{a}} \int \frac{d + e\,x^2}{\sqrt{1 + \frac{2\,c\,x^2}{b-q}}\,\sqrt{1 + \frac{2\,c\,x^2}{b+q}}}\,dx$$

- **Program code:**

```
Int[(d_.+e_.*x_^2)/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
  Module[{q=Rt[b^2-4*a*c,2]},
  Dist[1/Sqrt[a],Int[(d+e*x^2)/(Sqrt[1+2*c*x^2/(b-q)]*Sqrt[1+2*c*x^2/(b+q)]),x]]] /;
FreeQ[{a,b,c,d,e},x] && NonzeroQ[b^2-4*a*c] && PositiveQ[a]
```

- **Derivation: Algebraic expansion and piecewise constant extraction**
- **Basis: If** $q=\sqrt{b^2-4\,a\,c}$, **then** $a+b\,x^2+c\,x^4=a\left(1+\frac{2\,c\,x^2}{b-q}\right)\left(1+\frac{2\,c\,x^2}{b+q}\right)$
- **Basis: If** $q=\sqrt{b^2-4\,a\,c}$, **then** $\partial_x\frac{\sqrt{1+\frac{2\,c\,x^2}{b-q}}\,\sqrt{1+\frac{2\,c\,x^2}{b+q}}}{\sqrt{a+b\,x^2+c\,x^4}}=0$
- **Rule: If** $b^2-4\,a\,c\neq 0$, **let** $q=\sqrt{b^2-4\,a\,c}$, **then**

$$\int\frac{d+e\,x^2}{\sqrt{a+b\,x^2+c\,x^4}}\,dx\,\longrightarrow\,\frac{\sqrt{1+\frac{2\,c\,x^2}{b-q}}\,\sqrt{1+\frac{2\,c\,x^2}{b+q}}}{\sqrt{a+b\,x^2+c\,x^4}}\int\frac{d+e\,x^2}{\sqrt{1+\frac{2\,c\,x^2}{b-q}}\,\sqrt{1+\frac{2\,c\,x^2}{b+q}}}\,dx$$

- **Program code:**

```
Int[(d_.+e_.*x_^2)/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
  Module[{q=Rt[b^2-4*a*c,2]},
  Sqrt[1+2*c*x^2/(b-q)]*Sqrt[1+2*c*x^2/(b+q)]/Sqrt[a+b*x^2+c*x^4]*
  Int[(d+e*x^2)/(Sqrt[1+2*c*x^2/(b-q)]*Sqrt[1+2*c*x^2/(b+q)]),x]] /;
FreeQ[{a,b,c,d,e},x] && NonzeroQ[b^2-4*a*c]
```

$$\int \frac{1}{x \sqrt{a + b x^n + c x^{2n}}} \, dx$$

- **Reference: G&R 2.266.7, CRC 260**
- **Rule:**

$$\int \frac{1}{x \sqrt{b x^n + c x^{2n}}} \, dx \longrightarrow -\frac{2 \sqrt{b x^n + c x^{2n}}}{b n x^n}$$

- **Program code:**

```
Int[1/(x_*Sqrt[b_.*x_^n_.+c_.*x_^j_.]),x_Symbol] :=
  -2*Sqrt[b*x^n+c*x^j]/(b*n*x^n) /;
FreeQ[{b,c,n},x] && ZeroQ[j-2*n]
```

- **Reference: G&R 2.266.1, CRC 258**
- **Rule: If** $b^2 - 4 a c \neq 0 \wedge a > 0$, **then**

$$\int \frac{1}{x \sqrt{a + b x^n + c x^{2n}}} \, dx \longrightarrow -\frac{1}{\sqrt{a}\, n} \operatorname{ArcTanh}\left[\frac{2 a + b x^n}{2 \sqrt{a} \sqrt{a + b x^n + c x^{2n}}}\right]$$

- **Program code:**

```
Int[1/(x_*Sqrt[a_+b_.*x_^n_.+c_.*x_^j_.]),x_Symbol] :=
  -ArcTanh[(2*a+b*x^n)/(2*Rt[a,2]*Sqrt[a+b*x^n+c*x^j])]/(n*Rt[a,2]) /;
FreeQ[{a,b,c,n},x] && ZeroQ[j-2*n] && NonzeroQ[b^2-4*a*c] && PosQ[a]
```

- **Reference: G&R 2.266.3, CRC 259**
- **Rule: If** $b^2 - 4 a c \neq 0 \wedge \neg (a > 0)$, **then**

$$\int \frac{1}{x \sqrt{a + b x^n + c x^{2n}}} \, dx \longrightarrow \frac{1}{\sqrt{-a}\, n} \operatorname{ArcTan}\left[\frac{2 a + b x^n}{2 \sqrt{-a} \sqrt{a + b x^n + c x^{2n}}}\right]$$

- **Program code:**

```
Int[1/(x_*Sqrt[a_+b_.*x_^n_.+c_.*x_^j_.]),x_Symbol] :=
  ArcTan[(2*a+b*x^n)/(2*Rt[-a,2]*Sqrt[a+b*x^n+c*x^j])]/(n*Rt[-a,2]) /;
FreeQ[{a,b,c,n},x] && ZeroQ[j-2*n] && NonzeroQ[b^2-4*a*c] && NegQ[a]
```

$$\int \left(a + b\,x^n + c\,x^{2\,n}\right)^p\,dx$$

- **Derivation: Algebraic manipulation and piecewise constant extraction**
- **Basis: If** $p - \frac{1}{2} \in \mathbb{Z}$ **and** $b^2 - 4\,a\,c = 0$**, then** $\left(a + b\,x^n + c\,x^{2\,n}\right)^p = \frac{\sqrt{a+b\,x^n+c\,x^{2\,n}}}{(4\,c)^{p-\frac{1}{2}}\,(b+2\,c\,x^n)}\,(b + 2\,c\,x^n)^{2\,p}$
- **Basis: If** $b^2 - 4\,a\,c = 0$**, then** $\partial_x \frac{\sqrt{a+b\,x^n+c\,x^{2\,n}}}{(b+2\,c\,x^n)} = 0$
- **Rule: If** $n,\ p - \frac{1}{2} \in \mathbb{Z}\ \bigwedge\ n > 1\ \bigwedge\ p > 0\ \bigwedge\ b^2 - 4\,a\,c = 0$**, then**

$$\int \left(a + b\,x^n + c\,x^{2\,n}\right)^p\,dx \longrightarrow \frac{\sqrt{a + b\,x^n + c\,x^{2\,n}}}{(4\,c)^{p-\frac{1}{2}}\,(b + 2\,c\,x^n)} \int (b + 2\,c\,x^n)^{2\,p}\,dx$$

- **Program code:**

```
Int[(a_+b_.*x_^n_+c_.*x_^j_)^p_,x_Symbol] :=
  Sqrt[a+b*x^n+c*x^(2*n)]/(b+2*c*x^n)*Dist[1/(4*c)^(p-1/2),Int[(b+2*c*x^n)^(2*p),x]] /;
FreeQ[{a,b,c},x] && ZeroQ[j-2*n] && IntegersQ[n,p-1/2] && n>1 && p>0 && ZeroQ[b^2-4*a*c]
```

- **Derivation: Algebraic manipulation and piecewise constant extraction**
- **Basis: If** $p - \frac{1}{2} \in \mathbb{Z}$ **and** $b^2 - 4\,a\,c = 0$**, then** $\left(a + b\,x^n + c\,x^{2\,n}\right)^p = \frac{b+2\,c\,x^n}{(4\,c)^{p+\frac{1}{2}}\,\sqrt{a+b\,x^n+c\,x^{2\,n}}}\,(b + 2\,c\,x^n)^{2\,p}$
- **Basis: If** $b^2 - 4\,a\,c = 0$**, then** $\partial_x \frac{b+2\,c\,x^n}{\sqrt{a+b\,x^n+c\,x^{2\,n}}} = 0$
- **Rule: If** $n,\ p + \frac{1}{2} \in \mathbb{Z}\ \bigwedge\ n > 1\ \bigwedge\ p < 0\ \bigwedge\ b^2 - 4\,a\,c = 0$**, then**

$$\int \left(a + b\,x^n + c\,x^{2\,n}\right)^p\,dx \longrightarrow \frac{b + 2\,c\,x^n}{(4\,c)^{p+\frac{1}{2}}\,\sqrt{a + b\,x^n + c\,x^{2\,n}}} \int (b + 2\,c\,x^n)^{2\,p}\,dx$$

- **Program code:**

```
Int[(a_+b_.*x_^n_+c_.*x_^j_)^p_,x_Symbol] :=
  (b+2*c*x^n)/Sqrt[a+b*x^n+c*x^(2*n)]*Dist[1/(4*c)^(p+1/2),Int[(b+2*c*x^n)^(2*p),x]] /;
FreeQ[{a,b,c},x] && ZeroQ[j-2*n] && IntegersQ[n,p+1/2] && n>1 && p<0 && ZeroQ[b^2-4*a*c]
```

- **Note: Previously undiscovered rule?**
- **Note: Although the resulting integrand appears more complicated than the original, it has the form of another new rule.**
- **Rule: If** $n \in \mathbb{Z} \land n > 1 \land p \in \mathbb{F} \land p > 0 \land b^2 - 4 a c \neq 0 \land 2 n p + 1 \neq 0$**, then**

$$\int \left(a + b x^n + c x^{2 n}\right)^p dx \longrightarrow \frac{x \left(a + b x^n + c x^{2 n}\right)^p}{2 n p + 1} + \frac{n p}{2 n p + 1} \int (2 a + b x^n) \left(a + b x^n + c x^{2 n}\right)^{p-1} dx$$

- **Program code:**

```
Int[(a_+b_.*x_^n_+c_.*x_^j_)^p_,x_Symbol] :=
  x*(a+b*x^n+c*x^(2*n))^p/(2*n*p+1) +
  Dist[n*p/(2*n*p+1),Int[(2*a+b*x^n)*(a+b*x^n+c*x^(2*n))^(p-1),x]] /;
FreeQ[{a,b,c},x] && ZeroQ[j-2*n] && IntegerQ[n] && n>1 && FractionQ[p] && p>0 &&
NonzeroQ[b^2-4*a*c] && NonzeroQ[2*n*p+1]
```

$$\int x^m \left(a + b x^n + c x^{2n}\right)^p dx$$

- **Reference: G&R 2.265c**

- **Rule: If** $p \in \mathbb{F} \bigwedge p < -\frac{1}{2}$**, then**

$$\int \frac{\left(b x + c x^2\right)^p}{x} dx \longrightarrow \frac{\left(b x + c x^2\right)^{p+1}}{b p x} - \frac{c (2 p + 1)}{b p} \int \left(b x + c x^2\right)^p dx$$

- **Program code:**

```
Int[(b_.*x_+c_.*x_^2)^p_/x_,x_Symbol] :=
  (b*x+c*x^2)^(p+1)/(b*p*x) -
  Dist[c*(2*p+1)/(b*p),Int[(b*x+c*x^2)^p,x]] /;
FreeQ[{b,c},x] && FractionQ[p] && p<-1/2
```

- **Derivation: Algebraic manipulation and piecewise constant extraction**

- **Basis: If** $p - \frac{1}{2} \in \mathbb{Z}$ **and** $b^2 - 4 a c = 0$**, then** $(a + b x^n + c x^{2n})^p = \frac{\sqrt{a+b x^n+c x^{2n}}}{(4 c)^{p-\frac{1}{2}} (b+2 c x^n)} (b + 2 c x^n)^{2p}$

- **Basis: If** $b^2 - 4 a c = 0$**, then** $\partial_x \frac{\sqrt{a+b x^n+c x^{2n}}}{(b+2 c x^n)} = 0$

- **Rule: If** $m, n, p - \frac{1}{2} \in \mathbb{Z} \bigwedge n > 0 \bigwedge p > 0 \bigwedge b^2 - 4 a c = 0 \land m - n + 1 \neq 0$**, then**

$$\int x^m \left(a + b x^n + c x^{2n}\right)^p dx \longrightarrow \frac{\sqrt{a + b x^n + c x^{2n}}}{(4 c)^{p-\frac{1}{2}} (b + 2 c x^n)} \int x^m (b + 2 c x^n)^{2p} dx$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*x_^n_.+c_.*x_^j_)^p_,x_Symbol] :=
  Sqrt[a+b*x^n+c*x^(2*n)]/(b+2*c*x^n)*Dist[1/(4*c)^(p-1/2),Int[x^m*(b+2*c*x^n)^(2*p),x]] /;
FreeQ[{a,b,c},x] && ZeroQ[j-2*n] && IntegersQ[m,n,p-1/2] && n>0 && p>0 && ZeroQ[b^2-4*a*c] &&
NonzeroQ[m-n+1]
```

- **Derivation: Algebraic manipulation and piecewise constant extraction**
- **Basis: If** $p-\frac{1}{2}\in\mathbb{Z}$ **and** $b^2-4ac=0$**, then** $(a+bx^n+cx^{2n})^p=\frac{b+2cx^n}{(4c)^{p+\frac{1}{2}}\sqrt{a+bx^n+cx^{2n}}}(b+2cx^n)^{2p}$
- **Basis: If** $b^2-4ac=0$**, then** $\partial_x\frac{b+2cx^n}{\sqrt{a+bx^n+cx^{2n}}}=0$
- **Rule: If** $m,n,p+\frac{1}{2}\in\mathbb{Z}\bigwedge n>0\bigwedge p<0\bigwedge b^2-4ac=0\land m-n+1\neq 0$**, then**

$$\int x^m\left(a+bx^n+cx^{2n}\right)^p dx\longrightarrow\frac{b+2cx^n}{(4c)^{p+\frac{1}{2}}\sqrt{a+bx^n+cx^{2n}}}\int x^m(b+2cx^n)^{2p}dx$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*x_^n_.+c_.*x_^j_)^p_,x_Symbol] :=
  (b+2*c*x^n)/Sqrt[a+b*x^n+c*x^(2*n)]*Dist[1/(4*c)^(p+1/2),Int[x^m*(b+2*c*x^n)^(2*p),x]] /;
FreeQ[{a,b,c},x] && ZeroQ[j-2*n] && IntegersQ[m,n,p+1/2] && n>0 && p<0 && ZeroQ[b^2-4*a*c] &&
NonzeroQ[m-n+1]
```

- **Reference: G&R 2.174.2**
- **Note: Can this be generalized to handle any symmetric trinomial?**
- **Rule: If** $m\in\mathbb{Z}\bigwedge p\in\mathbb{F}\bigwedge p<-1\bigwedge m+2p+1=0$**, then**

$$\int x^m\left(a+bx+cx^2\right)^p dx\longrightarrow-\frac{x^{m-1}\left(a+bx+cx^2\right)^{p+1}}{c\,(m-1)}-\frac{b}{2c}\int x^{m-1}\left(a+bx+cx^2\right)^p dx+\frac{1}{c}\int x^{m-2}\left(a+bx+cx^2\right)^{p+1}dx$$

- **Program code:**

```
Int[x_^m_*(a_.+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  -x^(m-1)*(a+b*x+c*x^2)^(p+1)/(c*(m-1)) -
  Dist[b/(2*c),Int[x^(m-1)*(a+b*x+c*x^2)^p,x]] +
  Dist[1/c,Int[x^(m-2)*(a+b*x+c*x^2)^(p+1),x]] /;
FreeQ[{a,b,c},x] && IntegerQ[m] && FractionQ[p] && p<-1 && ZeroQ[m+2*p+1]
```

- **Reference: G&R 2.160.2**

- **Rule: If** $m, n \in \mathbb{Z} \wedge m < -1 \wedge n > 0 \wedge p \in \mathbb{F} \wedge p > 0$**, then**

$$\int x^m \left(a + b x^n + c x^{2n}\right)^p dx \longrightarrow \frac{x^{m+1} \left(a + b x^n + c x^{2n}\right)^p}{m+1} - \frac{b n p}{m+1} \int x^{m+n} \left(a + b x^n + c x^{2n}\right)^{p-1} dx - \frac{2 c n p}{m+1} \int x^{m+2n} \left(a + b x^n + c x^{2n}\right)^{p-1} dx$$

- **Program code:**

```
Int[x_^m_*(a_+b_.*x_^n_.+c_.*x_^j_)^p_,x_Symbol] :=
  x^(m+1)*(a+b*x^n+c*x^(2*n))^p/(m+1) -
  Dist[b*n*p/(m+1),Int[x^(m+n)*(a+b*x^n+c*x^(2*n))^(p-1),x]] -
  Dist[2*c*n*p/(m+1),Int[x^(m+2*n)*(a+b*x^n+c*x^(2*n))^(p-1),x]] /;
FreeQ[{a,b,c},x] && ZeroQ[j-2*n] && IntegersQ[m,n] && m<-1 && n>0 && FractionQ[p] && p>0
```

- **Reference: G&R 2.160.4**

- **Note: This rule is commented out since it seems inferior to G&R 2.160.2 above.**

- **Rule: If** $m, n \in \mathbb{Z} \wedge m < -1 \wedge n > 0 \wedge p \in \mathbb{F} \wedge p > 0 \wedge m + 2 n p + 1 \neq 0$**, then**

$$\int x^m \left(a + b x^n + c x^{2n}\right)^p dx \longrightarrow \frac{x^{m+1} \left(a + b x^n + c x^{2n}\right)^p}{m + 2 n p + 1} + \frac{2 a n p}{m + 2 n p + 1} \int x^m \left(a + b x^n + c x^{2n}\right)^{p-1} dx + \frac{b n p}{m + 2 n p + 1} \int x^{m+n} \left(a + b x^n + c x^{2n}\right)^{p-1} dx$$

- **Program code:**

```
(* Int[x_^m_*(a_+b_.*x_^n_.+c_.*x_^j_)^p_,x_Symbol] :=
  x^(m+1)*(a+b*x^n+c*x^(2*n))^p/(m+2*n*p+1) +
  Dist[2*a*n*p/(m+2*n*p+1),Int[x^m*(a+b*x^n+c*x^(2*n))^(p-1),x]] +
  Dist[b*n*p/(m+2*n*p+1),Int[x^(m+n)*(a+b*x^n+c*x^(2*n))^(p-1),x]] /;
FreeQ[{a,b,c},x] && ZeroQ[j-2*n] && IntegersQ[m,n] && m<-1 && n>0 && FractionQ[p] && p>0 &&
NonzeroQ[m+2*n*p+1] *)
```

- **Reference: G&R 2.160.1**

- **Note: G&R 2.161.6 is a special case of G&R 2.160.1.**

- **Rule: If** $m, n \in \mathbb{Z} \wedge m < -1 \wedge n > 0 \wedge p \in \mathbb{F} \wedge m + n (p + 1) + 1 \neq 0 \wedge m + 2 n (p + 1) + 1 \neq 0$**, then**

$$\int x^m \left(a + b x^n + c x^{2 n}\right)^p dx \longrightarrow \frac{x^{m+1} \left(a + b x^n + c x^{2 n}\right)^{p+1}}{a (m + 1)} - \frac{b (m + n (p + 1) + 1)}{a (m + 1)} \int x^{m+n} \left(a + b x^n + c x^{2 n}\right)^p dx - \frac{c (m + 2 n (p + 1) + 1)}{a (m + 1)} \int x^{m+2 n} \left(a + b x^n + c x^{2 n}\right)^p dx$$

- **Program code:**

```
Int[x_^m_*(a_+b_.*x_^n_.+c_.*x_^j_)^p_,x_Symbol] :=
  x^(m+1)*(a+b*x^n+c*x^(2*n))^(p+1)/(a*(m+1)) -
  Dist[b*(m+n*(p+1)+1)/(a*(m+1)),Int[x^(m+n)*(a+b*x^n+c*x^(2*n))^p,x]] -
  Dist[c*(m+2*n*(p+1)+1)/(a*(m+1)),Int[x^(m+2*n)*(a+b*x^n+c*x^(2*n))^p,x]] /;
FreeQ[{a,b,c,p},x] && ZeroQ[j-2*n] && IntegersQ[m,n] && m<-1 && n>0 && FractionQ[p] &&
NonzeroQ[m+n*(p+1)+1] && NonzeroQ[m+2*n*(p+1)+1]
```

- **Reference: G&R 2.160.3**

- **Rule: If** $m, n \in \mathbb{Z} \wedge 2 < 2 n \leq m \wedge p \in \mathbb{F} \wedge m + n (p - 1) + 1 = 0$**, then**

$$\int x^m \left(a + b x^n + c x^{2 n}\right)^p dx \longrightarrow \frac{x^{m-2 n+1} \left(a + b x^n + c x^{2 n}\right)^{p+1}}{c n (p + 1)} + \frac{a}{c} \int x^{m-2 n} \left(a + b x^n + c x^{2 n}\right)^p dx$$

- **Program code:**

```
Int[x_^m_*(a_+b_.*x_^n_.+c_.*x_^j_)^p_,x_Symbol] :=
  x^(m-2*n+1)*(a+b*x^n+c*x^(2*n))^(p+1)/(c*n*(p+1)) +
  Dist[a/c,Int[x^(m-2*n)*(a+b*x^n+c*x^(2*n))^p,x]] /;
FreeQ[{a,b,c,p},x] && ZeroQ[j-2*n] && IntegersQ[m,n] && 0<2*n≤m && FractionQ[p] &&
ZeroQ[m+n*(p-1)+1]
```

- **Reference: G&R 2.160.3**
- **Note: G&R 2.174.1 is a special case of G&R 2.160.3.**
- **Rule: If** $m, n \in \mathbb{Z} \land 2 < 2n \leq m \land p \in \mathbb{F} \land m + 2np + 1 \neq 0 \land m + n(p - 1) + 1 \neq 0$**, then**

$$\int x^m \left(a + b x^n + c x^{2n}\right)^p dx \longrightarrow \frac{x^{m-2n+1} \left(a + b x^n + c x^{2n}\right)^{p+1}}{c (m + 2np + 1)} - \frac{b (m + n (p - 1) + 1)}{c (m + 2np + 1)} \int x^{m-n} \left(a + b x^n + c x^{2n}\right)^p dx - \frac{a (m - 2n + 1)}{c (m + 2np + 1)} \int x^{m-2n} \left(a + b x^n + c x^{2n}\right)^p dx$$

- **Program code:**

```
Int[x_^m_*(a_+b_.*x_^n_.+c_.*x_^j_)^p_,x_Symbol] :=
  x^(m-2*n+1)*(a+b*x^n+c*x^(2*n))^(p+1)/(c*(m+2*n*p+1)) -
  Dist[b*(m+n*(p-1)+1)/(c*(m+2*n*p+1)),Int[x^(m-n)*(a+b*x^n+c*x^(2*n))^p,x]] -
  Dist[a*(m-2*n+1)/(c*(m+2*n*p+1)),Int[x^(m-2*n)*(a+b*x^n+c*x^(2*n))^p,x]] /;
FreeQ[{a,b,c,p},x] && ZeroQ[j-2*n] && IntegersQ[m,n] && 0<2*n≤m && FractionQ[p] &&
NonzeroQ[m+2*n*p+1] && NonzeroQ[m+n*(p-1)+1]
```

$$\int (d+e\,x^n)\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$$

- **Note: Previously undiscovered rule?**
- **Note: Although the resulting integrand has complicated coefficients, it is has the same form as the original integrand so recursion can occur.**
- **Rule: If** $n\in\mathbb{Z}\,\land\,n>1\,\land\,p\in\mathbb{F}\,\land\,p>0\,\land\,b^2-4\,a\,c\neq 0\,\land\,2\,n\,p+1\neq 0\,\land\,2\,n\,p+n+1\neq 0$**, then**

$$\int (d+e\,x^n)\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx\,\longrightarrow$$
$$\frac{x\,(b\,e\,n\,p+c\,d\,(2\,n\,p+n+1)+c\,e\,(2\,n\,p+1)\,x^n)}{c\,(2\,n\,p+1)\,(2\,n\,p+n+1)}\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p-\frac{n\,p}{c\,(2\,n\,p+1)\,(2\,n\,p+n+1)}\,\cdot$$
$$\int \left(a\,b\,e-2\,a\,c\,d\,(2\,n\,p+n+1)-\left(2\,a\,c\,e\,(2\,n\,p+1)+b\,c\,d\,(2\,n\,p+n+1)-b^2\,e\,(n\,p+1)\right)\,x^n\right)\,\cdot$$
$$\left(a+b\,x^n+c\,x^{2\,n}\right)^{p-1}\,dx$$

- **Program code:**

```
Int[(d_.+e_.*x_^n_)*(a_+b_.*x_^n_+c_.*x_^j_)^p_,x_Symbol] :=
  x*(b*e*n*p+c*d*(2*n*p+n+1)+c*e*(2*n*p+1)*x^n)*(a+b*x^n+c*x^(2*n))^p/(c*(2*n*p+1)*(2*n*p+n+1)) -
  Dist[n*p/(c*(2*n*p+1)*(2*n*p+n+1)),
    Int[(a*b*e-2*a*c*d*(2*n*p+n+1)-(2*a*c*e*(2*n*p+1)+b*c*d*(2*n*p+n+1)-b^2*e*(n*p+1))*x^n)*
      (a+b*x^n+c*x^(2*n))^(p-1),x]] /;
FreeQ[{a,b,c,d,e},x] && ZeroQ[j-2*n] && IntegerQ[n] && n>1 && FractionQ[p] && p>0 &&
NonzeroQ[b^2-4*a*c] && NonzeroQ[2*n*p+1] && NonzeroQ[2*n*p+n+1]
```

$\int \left(a + b\,x + c\,x^2 + d\,x^3 + e\,x^4\right)^p \,dx$

- **Derivation: Integration by substitution**

- **Basis: If** $d^3 - 4\,c\,d\,e + 8\,b\,e^2 = 0$ **and** $t = \frac{d}{4\,e} + x$, **then** $a + b\,x + c\,x^2 + d\,x^3 + e\,x^4 = a + \frac{5\,d^4}{256\,e^3} - \frac{c\,d^2}{16\,e^2} + \left(c - \frac{3\,d^2}{8\,e}\right) t^2 + e\,t^4$

- **Note: The substitution transforms a dense quartic polynomial into a symmetric quartic trinomial.**

- **Rule: If** $p - \frac{1}{2} \in \mathbb{Z}\ \bigwedge\ d^3 - 4\,c\,d\,e + 8\,b\,e^2 = 0$, **then**

$$\int \left(a + b\,x + c\,x^2 + d\,x^3 + e\,x^4\right)^p dx \longrightarrow \text{Subst}\left[\int \left(a + \frac{5\,d^4}{256\,e^3} - \frac{c\,d^2}{16\,e^2} + \left(c - \frac{3\,d^2}{8\,e}\right) x^2 + e\,x^4\right)^p dx,\ x,\ \frac{d}{4\,e} + x\right]$$

- **Program code:**

```
Int[(a_.+b_.*x_+c_.*x_^2+d_.*x_^3+e_.*x_^4)^p_,x_Symbol] :=
  Subst[Int[(a+5*d^4/(256*e^3)-c*d^2/(16*e^2)+(c-3*d^2/(8*e))*x^2+e*x^4)^p,x],x,d/(4*e)+x] /;
FreeQ[{a,b,c,d,e},x] && IntegerQ[p-1/2] && ZeroQ[d^3-4*c*d*e+8*b*e^2]
```

- **Derivation: Integration by substitution**

- **Basis: If** $b^3 - 4\,a\,b\,c + 8\,a^2\,d = 0$ **and** $t = \frac{b}{4\,a} + \frac{1}{x}$, **then**
$a + b\,x + c\,x^2 + d\,x^3 + e\,x^4 = \frac{a\left(-3\,b^4 + 16\,a\,b^2\,c - 64\,a^2\,b\,d + 256\,a^3\,e - 32\,a^2\left(3\,b^2 - 8\,a\,c\right) t^2 + 256\,a^4\,t^4\right)}{(b - 4\,a\,t)^4}$

- **Note: The substitution transforms a dense quartic polynomial into a symmetric quartic trinomial over the 4th power of a linear.**

- **Rule: If** $p - \frac{1}{2} \in \mathbb{Z}\ \bigwedge\ b^3 - 4\,a\,b\,c + 8\,a^2\,d = 0$, **then**

$$\int \left(a + b\,x + c\,x^2 + d\,x^3 + e\,x^4\right)^p dx \longrightarrow -16\,a^2$$

$$\text{Subst}\left[\int \left(\frac{a\left(-3\,b^4 + 16\,a\,b^2\,c - 64\,a^2\,b\,d + 256\,a^3\,e - 32\,a^2\left(3\,b^2 - 8\,a\,c\right) x^2 + 256\,a^4\,x^4\right)}{(b - 4\,a\,x)^4}\right)^p \bigg/ (b - 4\,a\,x)^2\,dx,\right.$$
$$\left. x,\ \frac{b}{4\,a} + \frac{1}{x}\right]$$

- **Program code:**

```
Int[(a_.+b_.*x_+c_.*x_^2+d_.*x_^3+e_.*x_^4)^p_,x_Symbol] :=
  Dist[-16*a^2,Subst[
    Int[(a*(-3*b^4+16*a*b^2*c-64*a^2*b*d+256*a^3*e-32*a^2*(3*b^2-8*a*c)*x^2+256*a^4*x^4)/(b-4*a*x)^4
FreeQ[{a,b,c,d,e},x] && IntegerQ[p-1/2] && ZeroQ[b^3-4*a*b*c+8*a^2*d]
```

```
Int[u_^p_,x_Symbol] :=
  Module[{v=Expand[u,x]},
  Int[v^p,x] /;
 v=!=u && (
 MatchQ[v,a_+b_.*x^2+c_.*x^4 /; FreeQ[{a,b,c},x]] ||
 MatchQ[v,a_.+b_.*x+c_.*x^2+d_.*x^3+e_.*x^4 /; FreeQ[{a,b,c,d,e},x] && ZeroQ[d^3-4*c*d*e+8*b*e^2]])]
PolynomialQ[u,x] && Exponent[u,x]==4 && IntegerQ[p-1/2]
```