$$\int \mathrm{ArcSinh}[a + b\,x]^n \, dx$$

- **Reference: CRC 579, A&S 4.6.43**
- **Derivation: Integration by parts**
- **Rule:**

$$\int \mathrm{ArcSinh}[a + b\,x] \, dx \longrightarrow \frac{(a + b\,x)\ \mathrm{ArcSinh}[a + b\,x]}{b} - \frac{\sqrt{1 + (a + b\,x)^2}}{b}$$

- **Program code:**

```
Int[ArcSinh[a_.+b_.*x_],x_Symbol] :=
  (a+b*x)*ArcSinh[a+b*x]/b - Sqrt[1+(a+b*x)^2]/b /;
FreeQ[{a,b},x]
```

- **Derivation: Iterated integration by parts**
- **Rule: If $n > 1$, then**

$$\int \mathrm{ArcSinh}[a + b\,x]^n \, dx \longrightarrow \frac{(a + b\,x)\ \mathrm{ArcSinh}[a + b\,x]^n}{b} - \frac{n\sqrt{1 + (a + b\,x)^2}\ \mathrm{ArcSinh}[a + b\,x]^{n-1}}{b} + n\,(n-1)\int \mathrm{ArcSinh}[a + b\,x]^{n-2}\, dx$$

- **Program code:**

```
Int[ArcSinh[a_.+b_.*x_]^n_,x_Symbol] :=
  (a+b*x)*ArcSinh[a+b*x]^n/b -
  n*Sqrt[1+(a+b*x)^2]*ArcSinh[a+b*x]^(n-1)/b +
  Dist[n*(n-1),Int[ArcSinh[a+b*x]^(n-2),x]] /;
FreeQ[{a,b},x] && RationalQ[n] && n>1
```

- **Derivation: Integration by substitution**
- **Basis:** $\frac{1}{\text{ArcSinh}[z]} = \frac{\text{Cosh}[\text{ArcSinh}[z]]}{\text{ArcSinh}[z]} \text{ArcSinh}'[z]$
- **Rule:**

$$\int \frac{1}{\text{ArcSinh}[a+b\,x]}\,dx \longrightarrow \frac{\text{CoshIntegral}[\text{ArcSinh}[a+b\,x]]}{b}$$

- **Program code:**

```
Int[1/ArcSinh[a_.+b_.*x_],x_Symbol] :=
  CoshIntegral[ArcSinh[a+b*x]]/b /;
FreeQ[{a,b},x]
```

- **Derivation: Integration by substitution**
- **Basis:** $\frac{1}{\text{ArcSinh}[z]^2} = \frac{\text{Cosh}[\text{ArcSinh}[z]]}{\text{ArcSinh}[z]^2} \text{ArcSinh}'[z]$
- **Rule:**

$$\int \frac{1}{\text{ArcSinh}[a+b\,x]^2}\,dx \longrightarrow -\frac{\sqrt{1+(a+b\,x)^2}}{b\,\text{ArcSinh}[a+b\,x]} + \frac{\text{SinhIntegral}[\text{ArcSinh}[a+b\,x]]}{b}$$

- **Program code:**

```
Int[1/ArcSinh[a_.+b_.*x_]^2,x_Symbol] :=
  -Sqrt[1+(a+b*x)^2]/(b*ArcSinh[a+b*x]) + SinhIntegral[ArcSinh[a+b*x]]/b /;
FreeQ[{a,b},x]
```

- **Derivation: Integration by substitution**
- **Basis:** $\frac{1}{\sqrt{\text{ArcSinh}[z]}} = \frac{\text{Cosh}[\text{ArcSinh}[z]]}{\sqrt{\text{ArcSinh}[z]}} \text{ArcSinh}'[z]$
- **Rule:**

$$\int \frac{1}{\sqrt{\text{ArcSinh}[a+b\,x]}}\,dx \longrightarrow \frac{\sqrt{\pi}\ \text{Erf}\left[\sqrt{\text{ArcSinh}[a+b\,x]}\right]}{2\,b} + \frac{\sqrt{\pi}\ \text{Erfi}\left[\sqrt{\text{ArcSinh}[a+b\,x]}\right]}{2\,b}$$

- **Program code:**

```
Int[1/Sqrt[ArcSinh[a_.+b_.*x_]],x_Symbol] :=
  Sqrt[Pi]*Erf[Sqrt[ArcSinh[a+b*x]]]/(2*b) +
  Sqrt[Pi]*Erfi[Sqrt[ArcSinh[a+b*x]]]/(2*b) /;
FreeQ[{a,b},x]
```

- **Derivation: Integration by parts**
- **Rule:**

$$\int\sqrt{\mathrm{ArcSinh}[a+b\,x]}\,dx \longrightarrow \frac{(a+b\,x)\sqrt{\mathrm{ArcSinh}[a+b\,x]}}{b}+\frac{\sqrt{\pi}\,\mathrm{Erf}\left[\sqrt{\mathrm{ArcSinh}[a+b\,x]}\right]}{4\,b}-\frac{\sqrt{\pi}\,\mathrm{Erfi}\left[\sqrt{\mathrm{ArcSinh}[a+b\,x]}\right]}{4\,b}$$

- **Program code:**

```
Int[Sqrt[ArcSinh[a_.+b_.*x_]],x_Symbol] :=
  (a+b*x)*Sqrt[ArcSinh[a+b*x]]/b +
  Sqrt[Pi]*Erf[Sqrt[ArcSinh[a+b*x]]]/(4*b) -
  Sqrt[Pi]*Erfi[Sqrt[ArcSinh[a+b*x]]]/(4*b) /;
FreeQ[{a,b},x]
```

- **Derivation: Inverted iterated integration by parts**
- **Rule: If $n<-1\wedge n\neq-2$, then**

$$\int\mathrm{ArcSinh}[a+b\,x]^n\,dx \longrightarrow -\frac{(a+b\,x)\,\mathrm{ArcSinh}[a+b\,x]^{n+2}}{b\,(n+1)\,(n+2)}+\frac{\sqrt{1+(a+b\,x)^2}\,\mathrm{ArcSinh}[a+b\,x]^{n+1}}{b\,(n+1)}+\frac{1}{(n+1)\,(n+2)}\int\mathrm{ArcSinh}[a+b\,x]^{n+2}\,dx$$

- **Program code:**

```
Int[ArcSinh[a_.+b_.*x_]^n_,x_Symbol] :=
  -(a+b*x)*ArcSinh[a+b*x]^(n+2)/(b*(n+1)*(n+2)) +
  Sqrt[1+(a+b*x)^2]*ArcSinh[a+b*x]^(n+1)/(b*(n+1)) +
  Dist[1/((n+1)*(n+2)),Int[ArcSinh[a+b*x]^(n+2),x]] /;
FreeQ[{a,b},x] && RationalQ[n] && n<-1 && n!=-2
```

- **Rule: If** $n \notin \mathbb{Q} \lor -1 < n < 1$**, then**

$$\int \text{ArcSinh}[a+b\,x]^n\,dx \longrightarrow \frac{\text{ArcSinh}[a+b\,x]^n\,\text{Gamma}[n+1,\,-\text{ArcSinh}[a+b\,x]]}{2\,b\,(-\text{ArcSinh}[a+b\,x])^n} - \frac{\text{Gamma}[n+1,\,\text{ArcSinh}[a+b\,x]]}{2\,b}$$

- **Program code:**

```
Int[ArcSinh[a_.+b_.*x_]^n_,x_Symbol] :=
  ArcSinh[a+b*x]^n*Gamma[n+1,-ArcSinh[a+b*x]]/(2*b*(-ArcSinh[a+b*x])^n) -
  Gamma[n+1,ArcSinh[a+b*x]]/(2*b) /;
FreeQ[{a,b,n},x] && (Not[RationalQ[n]] || -1<n<1)
```

$$\int x^m \operatorname{ArcSinh}[a+b\,x]\,dx$$

- **Reference: CRC 581, A&S 4.6.50**
- **Derivation: Integration by parts**
- **Rule: If $m+1 \neq 0$, then**

$$\int x^m \operatorname{ArcSinh}[a+b\,x]\,dx \longrightarrow \frac{x^{m+1}\operatorname{ArcSinh}[a+b\,x]}{m+1} - \frac{b}{m+1}\int \frac{x^{m+1}}{\sqrt{1+a^2+2\,a\,b\,x+b^2\,x^2}}\,dx$$

- **Program code:**

```
Int[x_^m_.*ArcSinh[a_.+b_.*x_],x_Symbol] :=
  x^(m+1)*ArcSinh[a+b*x]/(m+1) -
  Dist[b/(m+1),Int[x^(m+1)/Sqrt[1+a^2+2*a*b*x+b^2*x^2],x]] /;
FreeQ[{a,b,m},x] && NonzeroQ[m+1]
```

$$\int x^m \text{ArcSinh}[a\,x]^n \, dx$$

- **Rule:**

$$\int \frac{x}{\sqrt{\text{ArcSinh}[a\,x]}} \, dx \longrightarrow -\frac{1}{4\,a^2} \sqrt{\frac{\pi}{2}} \, \text{Erf}\left[\sqrt{2} \sqrt{\text{ArcSinh}[a\,x]}\right] + \frac{1}{4\,a^2} \sqrt{\frac{\pi}{2}} \, \text{Erfi}\left[\sqrt{2} \sqrt{\text{ArcSinh}[a\,x]}\right]$$

- **Program code:**

```
Int[x_/Sqrt[ArcSinh[a_.*x_]],x_Symbol] :=
  -Sqrt[Pi/2]*Erf[Sqrt[2]*Sqrt[ArcSinh[a*x]]]/(4*a^2) +
  Sqrt[Pi/2]*Erfi[Sqrt[2]*Sqrt[ArcSinh[a*x]]]/(4*a^2) /;
FreeQ[a,x]
```

- **Rule:**

$$\int \frac{x}{\text{ArcSinh}[a\,x]^{3/2}} \, dx \longrightarrow$$

$$-\frac{2\,x\,\sqrt{1+a^2\,x^2}}{a\,\sqrt{\text{ArcSinh}[a\,x]}} + \frac{1}{a^2} \sqrt{\frac{\pi}{2}} \, \text{Erf}\left[\sqrt{2} \sqrt{\text{ArcSinh}[a\,x]}\right] + \frac{1}{a^2} \sqrt{\frac{\pi}{2}} \, \text{Erfi}\left[\sqrt{2} \sqrt{\text{ArcSinh}[a\,x]}\right]$$

- **Program code:**

```
Int[x_/ArcSinh[a_.*x_]^(3/2),x_Symbol] :=
  -2*x*Sqrt[1+a^2*x^2]/(a*Sqrt[ArcSinh[a*x]]) +
  Sqrt[Pi/2]*Erf[Sqrt[2]*Sqrt[ArcSinh[a*x]]]/a^2 +
  Sqrt[Pi/2]*Erfi[Sqrt[2]*Sqrt[ArcSinh[a*x]]]/a^2 /;
FreeQ[a,x]
```

- **Rule: If $n > 1$, then**

$$\int x\,\text{ArcSinh}[a\,x]^n\,dx \longrightarrow -\frac{n\,x\,\sqrt{1+a^2\,x^2}\,\text{ArcSinh}[a\,x]^{n-1}}{4\,a}+\frac{\text{ArcSinh}[a\,x]^n}{4\,a^2}+\frac{x^2\,\text{ArcSinh}[a\,x]^n}{2}+\frac{n\,(n-1)}{4}\int x\,\text{ArcSinh}[a\,x]^{n-2}\,dx$$

- **Program code:**

```
Int[x_*ArcSinh[a_.*x_]^n_,x_Symbol] :=
  -n*x*Sqrt[1+a^2*x^2]*ArcSinh[a*x]^(n-1)/(4*a) +
  ArcSinh[a*x]^n/(4*a^2) + x^2*ArcSinh[a*x]^n/2 +
  Dist[n*(n-1)/4,Int[x*ArcSinh[a*x]^(n-2),x]] /;
FreeQ[a,x] && RationalQ[n] && n>0
```

- **Rule: If $n < -1 \land n \neq -2$, then**

$$\int x\,\text{ArcSinh}[a\,x]^n\,dx \longrightarrow \frac{x\,\sqrt{1+a^2\,x^2}\,\text{ArcSinh}[a\,x]^{n+1}}{a\,(n+1)}-\frac{\text{ArcSinh}[a\,x]^{n+2}}{a^2\,(n+1)\,(n+2)}-\frac{2\,x^2\,\text{ArcSinh}[a\,x]^{n+2}}{(n+1)\,(n+2)}+\frac{4}{(n+1)\,(n+2)}\int x\,\text{ArcSinh}[a\,x]^{n+2}\,dx$$

- **Program code:**

```
Int[x_*ArcSinh[a_.*x_]^n_,x_Symbol] :=
  x*Sqrt[1+a^2*x^2]*ArcSinh[a*x]^(n+1)/(a*(n+1)) -
  ArcSinh[a*x]^(n+2)/(a^2*(n+1)*(n+2)) -
  2*x^2*ArcSinh[a*x]^(n+2)/((n+1)*(n+2)) +
  Dist[4/((n+1)*(n+2)),Int[x*ArcSinh[a*x]^(n+2),x]] /;
FreeQ[a,x] && RationalQ[n] && n<-1 && n≠-2
```

- **Rule: If $n > 1$, then**

$$\int \frac{\text{ArcSinh}[a\,x]^n}{x^3}\,dx \longrightarrow -\frac{a\,n\,\sqrt{1+a^2\,x^2}\,\text{ArcSinh}[a\,x]^{n-1}}{2\,x}-\frac{\text{ArcSinh}[a\,x]^n}{2\,x^2}+\frac{a^2\,n\,(n-1)}{2}\int \frac{\text{ArcSinh}[a\,x]^{n-2}}{x}\,dx$$

- **Program code:**

```
Int[ArcSinh[a_.*x_]^n_/x_^3,x_Symbol] :=
  -a*n*Sqrt[1+a^2*x^2]*ArcSinh[a*x]^(n-1)/(2*x) -
  ArcSinh[a*x]^n/(2*x^2) +
  Dist[a^2*n*(n-1)/2,Int[ArcSinh[a*x]^(n-2)/x,x]] /;
FreeQ[a,x] && RationalQ[n] && n>1
```

- **Rule: If $m \in \mathbb{Z} \land m < -3 \land n > 1$, then**

$$\int x^m\,\text{ArcSinh}[a\,x]^n\,dx \longrightarrow -\frac{a\,n\,x^{m+2}\,\sqrt{1+a^2\,x^2}\,\text{ArcSinh}[a\,x]^{n-1}}{(m+1)\,(m+2)}+\frac{x^{m+1}\,\text{ArcSinh}[a\,x]^n}{(m+1)}+\frac{a^2\,(m+3)\,x^{m+3}\,\text{ArcSinh}[a\,x]^n}{(m+1)\,(m+2)}-\frac{a^2\,(m+3)^2}{(m+1)\,(m+2)}\int x^{m+2}\,\text{ArcSinh}[a\,x]^n\,dx+\frac{a^2\,n\,(n-1)}{(m+1)\,(m+2)}\int x^{m+2}\,\text{ArcSinh}[a\,x]^{n-2}\,dx$$

- **Program code:**

```
Int[x_^m_*ArcSinh[a_.*x_]^n_,x_Symbol] :=
  -a*n*x^(m+2)*Sqrt[1+a^2*x^2]*ArcSinh[a*x]^(n-1)/((m+1)*(m+2)) +
  x^(m+1)*ArcSinh[a*x]^n/(m+1) +
  a^2*(m+3)*x^(m+3)*ArcSinh[a*x]^n/((m+1)*(m+2)) -
  Dist[a^2*(m+3)^2/((m+1)*(m+2)),Int[x^(m+2)*ArcSinh[a*x]^n,x]] +
  Dist[a^2*n*(n-1)/((m+1)*(m+2)),Int[x^(m+2)*ArcSinh[a*x]^(n-2),x]] /;
FreeQ[a,x] && IntegerQ[m] && RationalQ[n] && m<-3 && n>1
```

- **Rule: If** $m \in \mathbb{Z} \wedge m > 1 \wedge n < -1 \wedge n \neq -2$**, then**

$$\int x^m \operatorname{ArcSinh}[a x]^n \, dx \longrightarrow \frac{x^m \sqrt{1 + a^2 x^2} \operatorname{ArcSinh}[a x]^{n+1}}{a (n+1)} - \frac{m x^{m-1} \operatorname{ArcSinh}[a x]^{n+2}}{a^2 (n+1)(n+2)} - \frac{(m+1) x^{m+1} \operatorname{ArcSinh}[a x]^{n+2}}{(n+1)(n+2)} + \frac{(m+1)^2}{(n+1)(n+2)} \int x^m \operatorname{ArcSinh}[a x]^{n+2} \, dx + \frac{m (m-1)}{a^2 (n+1)(n+2)} \int x^{m-2} \operatorname{ArcSinh}[a x]^{n+2} \, dx$$

- **Program code:**

```
Int[x_^m_*ArcSinh[a_.*x_]^n_,x_Symbol] :=
  x^m*Sqrt[1+a^2*x^2]*ArcSinh[a*x]^(n+1)/(a*(n+1)) -
  m*x^(m-1)*ArcSinh[a*x]^(n+2)/(a^2*(n+1)*(n+2)) -
  (m+1)*x^(m+1)*ArcSinh[a*x]^(n+2)/((n+1)*(n+2)) +
  Dist[(m+1)^2/((n+1)*(n+2)),Int[x^m*ArcSinh[a*x]^(n+2),x]] +
  Dist[m*(m-1)/(a^2*(n+1)*(n+2)),Int[x^(m-2)*ArcSinh[a*x]^(n+2),x]] /;
FreeQ[a,x] && IntegerQ[m] && RationalQ[n] && m>1 && n<-1 && n≠-2
```

- **Derivation: Integration by substitution**

- **Basis:** $\frac{\operatorname{ArcSinh}[a x^p]^n}{x} = \frac{1}{p} \operatorname{ArcSinh}[a x^p]^n \operatorname{Coth}[\operatorname{ArcSinh}[a x^p]] \, \partial_x \operatorname{ArcSinh}[a x^p]$

- **Rule: If** $n \in \mathbb{Z} \wedge n > 0$**, then**

$$\int \frac{\operatorname{ArcSinh}[a x^p]^n}{x} \, dx \longrightarrow \frac{1}{p} \operatorname{Subst}\left[\int x^n \operatorname{Coth}[x] \, dx, x, \operatorname{ArcSinh}[a x^p]\right]$$

- **Program code:**

```
Int[ArcSinh[a_.*x_^p_.]^n_./x_,x_Symbol] :=
  Dist[1/p,Subst[Int[x^n*Coth[x],x],x,ArcSinh[a*x^p]]] /;
FreeQ[{a,p},x] && IntegerQ[n] && n>0
```

- **Derivation: Integration by parts and substitution**

- **Basis: If** $m \in \mathbb{Z}$, $\frac{x^{m+1} \text{ArcSinh}[a\,x]^{n-1}}{\sqrt{1+a^2 x^2}} = \frac{1}{a^{m+2}} \text{ArcSinh}[a\,x]^{n-1} \text{Sinh}[\text{ArcSinh}[a\,x]]^{m+1} \partial_x \text{ArcSinh}[a\,x]$

- **Rule: If** $m \in \mathbb{Z} \wedge m \neq -1$, **then**

$$\int x^m \text{ArcSinh}[a\,x]^n \, dx \longrightarrow \frac{x^{m+1} \text{ArcSinh}[a\,x]^n}{m+1} - \frac{n}{a^{m+1}\,(m+1)} \text{Subst}\left[\int x^{n-1} \text{Sinh}[x]^{m+1} \, dx,\ x,\ \text{ArcSinh}[a\,x]\right]$$

- **Program code:**

```
Int[x_^m_.*ArcSinh[a_.*x_]^n_,x_Symbol] :=
  x^(m+1)*ArcSinh[a*x]^n/(m+1) -
  Dist[n/(a^(m+1)*(m+1)),Subst[Int[x^(n-1)*Sinh[x]^(m+1),x],x,ArcSinh[a*x]]] /;
FreeQ[{a,n},x] && IntegerQ[m] && m≠-1
```

$$\int (a+b\,\text{ArcSinh}[c+d\,x])^n\,dx$$

- **Derivation: Integration by substitution**

- **Basis:** $(a+b\,\text{ArcSinh}[c+d\,x])^n = \frac{1}{d}\,(a+b\,\text{ArcSinh}[c+d\,x])^n\,\text{Cosh}[\text{ArcSinh}[c+d\,x]]\,\partial_x\text{ArcSinh}[c+d\,x]$

- **Rule:** If $n \notin \mathbb{Z}$, then

$$\int (a+b\,\text{ArcSinh}[c+d\,x])^n\,dx \longrightarrow \frac{1}{d}\,\text{Subst}\Big[\int (a+b\,x)^n\,\text{Cosh}[x]\,dx,\ x,\ \text{ArcSinh}[c+d\,x]\Big]$$

- **Program code:**

```
Int[(a_+b_.*ArcSinh[c_.+d_.*x_])^n_,x_Symbol] :=
  Dist[1/d,Subst[Int[(a+b*x)^n*Cosh[x],x],x,ArcSinh[c+d*x]]] /;
FreeQ[{a,b,c,d},x] && Not[IntegerQ[n]]
```

$$\int x^m \,(a + b\,\mathrm{ArcSinh}[c + d\,x])^n\,dx$$

- **Derivation: Integration by substitution**
- **Basis: If** $m \in \mathbb{Z}$, $x^m\,(a + b\,\mathrm{ArcSinh}[c + d\,x])^n =$

$$\frac{1}{d^{m+1}}\,(a + b\,\mathrm{ArcSinh}[c + d\,x])^n\,(\mathrm{Sinh}[\mathrm{ArcSinh}[c + d\,x]] - c)^m\,\mathrm{Cosh}[\mathrm{ArcSinh}[c + d\,x]]\,\partial_x \mathrm{ArcSinh}[c + d\,x]$$

- **Rule: If** $m \in \mathbb{Z} \wedge n \notin \mathbb{Z} \wedge m > 0$**, then**

$$\int x^m\,(a + b\,\mathrm{ArcSinh}[c + d\,x])^n\,dx \longrightarrow \frac{1}{d^{m+1}}\,\mathrm{Subst}\Big[\int (a + b\,x)^n\,(\mathrm{Sinh}[x] - c)^m\,\mathrm{Cosh}[x]\,dx,\ x,\ \mathrm{ArcSinh}[c + d\,x]\Big]$$

- **Program code:**

```
Int[x_^m_.*(a_.+b_.*ArcSinh[c_.+d_.*x_])^n_,x_Symbol] :=
  Dist[1/d^(m+1),Subst[Int[(a+b*x)^n*(Sinh[x]-c)^m*Cosh[x],x],x,ArcSinh[c+d*x]]] /;
FreeQ[{a,b,c,d},x] && IntegerQ[m] && Not[IntegerQ[n]] && m>0
```

$$\int \frac{x \operatorname{ArcSinh}[a+b\,x]^n}{\sqrt{1+(a+b\,x)^2}}\,dx$$

- **Derivation: Integration by parts**
- **Rule: If $n>1$, then**

$$\int \frac{x \operatorname{ArcSinh}[a+b\,x]^n}{\sqrt{1+(a+b\,x)^2}}\,dx \longrightarrow$$

$$\frac{\sqrt{1+(a+b\,x)^2}\;\operatorname{ArcSinh}[a+b\,x]^n}{b^2}-\frac{n}{b}\int \operatorname{ArcSinh}[a+b\,x]^{n-1}\,dx-\frac{a}{b}\int \frac{\operatorname{ArcSinh}[a+b\,x]^n}{\sqrt{1+(a+b\,x)^2}}\,dx$$

- **Program code:**

```
Int[x_*ArcSinh[a_.+b_.*x_]^n_/Sqrt[u_],x_Symbol] :=
  Sqrt[u]*ArcSinh[a+b*x]^n/b^2 -
  Dist[n/b,Int[ArcSinh[a+b*x]^(n-1),x]] -
  Dist[a/b,Int[ArcSinh[a+b*x]^n/Sqrt[u],x]] /;
FreeQ[{a,b},x] && ZeroQ[u-1-(a+b*x)^2] && RationalQ[n] && n>1
```

$$\int u\,\text{ArcSinh}\left[\frac{c}{a+b\,x^n}\right]^m dx$$

- **Derivation: Algebraic simplification**

- **Basis:** $\text{ArcSinh}[z] = \text{ArcCsch}\left[\frac{1}{z}\right]$

- **Rule:**

$$\int u\,\text{ArcSinh}\left[\frac{c}{a+b\,x^n}\right]^m dx \longrightarrow \int u\,\text{ArcCsch}\left[\frac{a}{c}+\frac{b\,x^n}{c}\right]^m dx$$

- **Program code:**

```
Int[u_.*ArcSinh[c_./(a_.+b_.*x_^n_.)]^m_.,x_Symbol] :=
  Int[u*ArcCsch[a/c+b*x^n/c]^m,x] /;
FreeQ[{a,b,c,n,m},x]
```

$$\int f^{c\,\text{ArcSinh}[a+b\,x]}\,dx$$

- **Rule: If** $1-c^2\,\text{Log}[f]^2\neq 0$**, then**

$$\int f^{c\,\text{ArcSinh}[a+b\,x]}\,dx\;\longrightarrow\;\frac{a+b\,x-c\,\sqrt{1+(a+b\,x)^2}\;\text{Log}[f]}{b\,\left(1-c^2\,\text{Log}[f]^2\right)}\,f^{c\,\text{ArcSinh}[a+b\,x]}$$

- **Program code:**

```
Int[f_^(c_.*ArcSinh[a_.+b_.*x_]),x_Symbol] :=
  f^(c*ArcSinh[a+b*x])*(a+b*x-c*Sqrt[1+(a+b*x)^2]*Log[f])/(b*(1-c^2*Log[f]^2)) /;
FreeQ[{a,b,c,f},x] && NonzeroQ[1-c^2*Log[f]^2]
```

$$\int \mathrm{ArcSinh}[u]\, dx$$

- **Derivation: Integration by parts**
- **Rule: If u is free of inverse functions, then**

$$\int \mathrm{ArcSinh}[u]\, dx \longrightarrow x\, \mathrm{ArcSinh}[u] - \int \frac{x\, \partial_x u}{\sqrt{1+u^2}}\, dx$$

- **Program code:**

```
Int[ArcSinh[u_],x_Symbol] :=
  x*ArcSinh[u] -
  Int[Regularize[x*D[u,x]/Sqrt[1+u^2],x],x] /;
InverseFunctionFreeQ[u,x] && Not[FunctionOfExponentialOfLinear[u,x]]
```

$$\int x^m e^{n \operatorname{ArcSinh}[u]} \, dx$$

- **Derivation: Algebraic simplification**
- **Basis:** $e^{n \operatorname{ArcSinh}[z]} = \left(z + \sqrt{1 + z^2}\right)^n$
- **Rule: If** $n \in \mathbb{Z} \land u$ **is a polynomial in** x**, then**

$$\int e^{n \operatorname{ArcSinh}[u]} \, dx \longrightarrow \int \left(u + \sqrt{1 + u^2}\right)^n dx$$

- **Program code:**

```
Int[E^(n_.*ArcSinh[u_]), x_Symbol] :=
  Int[(u+Sqrt[1+u^2])^n,x] /;
IntegerQ[n] && PolynomialQ[u,x]
```

- **Derivation: Algebraic simplification**
- **Basis:** $e^{n \operatorname{ArcSinh}[z]} = \left(z + \sqrt{1 + z^2}\right)^n$
- **Rule: If** $n \in \mathbb{Z} \land u$ **is a polynomial in** x**, then**

$$\int x^m e^{n \operatorname{ArcSinh}[u]} \, dx \longrightarrow \int x^m \left(u + \sqrt{1 + u^2}\right)^n dx$$

- **Program code:**

```
Int[x_^m_.*E^(n_.*ArcSinh[u_]), x_Symbol] :=
  Int[x^m*(u+Sqrt[1+u^2])^n,x] /;
RationalQ[m] && IntegerQ[n] && PolynomialQ[u,x]
```