$\int \text{ArcCosh}[a+b\,x]^n\,dx$

- **Reference: CRC 582', A&S 4.6.44**
- **Derivation: Integration by parts**
- **Rule:**

$$\int \text{ArcCosh}[a+b\,x]\,dx \longrightarrow \frac{(a+b\,x)\,\text{ArcCosh}[a+b\,x]}{b}-\frac{\sqrt{-1+a+b\,x}\,\sqrt{1+a+b\,x}}{b}$$

- **Program code:**

```
Int[ArcCosh[a_.+b_.*x_],x_Symbol] :=
  (a+b*x)*ArcCosh[a+b*x]/b - Sqrt[-1+a+b*x]*Sqrt[1+a+b*x]/b /;
FreeQ[{a,b},x]
```

- **Derivation: Iterated integration by parts**
- **Rule: If $n>1$, then**

$$\int \text{ArcCosh}[a+b\,x]^n\,dx \longrightarrow \frac{(a+b\,x)\,\text{ArcCosh}[a+b\,x]^n}{b}-\frac{n\,\sqrt{-1+a+b\,x}\,\sqrt{1+a+b\,x}\,\text{ArcCosh}[a+b\,x]^{n-1}}{b}+n\,(n-1)\int \text{ArcCosh}[a+b\,x]^{n-2}\,dx$$

- **Program code:**

```
Int[ArcCosh[a_.+b_.*x_]^n_,x_Symbol] :=
  (a+b*x)*ArcCosh[a+b*x]^n/b -
  n*Sqrt[-1+a+b*x]*Sqrt[1+a+b*x]*ArcCosh[a+b*x]^(n-1)/b +
  Dist[n*(n-1),Int[ArcCosh[a+b*x]^(n-2),x]] /;
FreeQ[{a,b},x] && RationalQ[n] && n>1
```

- **Derivation: Integration by substitution**
- **Basis:** $\frac{1}{\text{ArcCosh}[z]} = \frac{\text{Sinh}[\text{ArcCosh}[z]]}{\text{ArcCosh}[z]} \text{ArcCosh}'[z]$
- **Rule:**

$$\int \frac{1}{\text{ArcCosh}[a+b\,x]}\,dx \longrightarrow \frac{\text{SinhIntegral}[\text{ArcCosh}[a+b\,x]]}{b}$$

- **Program code:**

```
Int[1/ArcCosh[a_.+b_.*x_],x_Symbol] :=
  SinhIntegral[ArcCosh[a+b*x]]/b /;
FreeQ[{a,b},x]
```

- **Derivation: Integration by substitution**
- **Basis:** $\frac{1}{\text{ArcCosh}[z]^2} = \frac{\text{Sinh}[\text{ArcCosh}[z]]}{\text{ArcCosh}[z]^2} \text{ArcCosh}'[z]$
- **Rule:**

$$\int \frac{1}{\text{ArcCosh}[a+b\,x]^2}\,dx \longrightarrow -\frac{\sqrt{-1+a+b\,x}\,\sqrt{1+a+b\,x}}{b\,\text{ArcCosh}[a+b\,x]} + \frac{\text{CoshIntegral}[\text{ArcCosh}[a+b\,x]]}{b}$$

- **Program code:**

```
Int[1/ArcCosh[a_.+b_.*x_]^2,x_Symbol] :=
  -Sqrt[-1+a+b*x]*Sqrt[1+a+b*x]/(b*ArcCosh[a+b*x]) + CoshIntegral[ArcCosh[a+b*x]]/b /;
FreeQ[{a,b},x]
```

- **Derivation: Integration by substitution**
- **Basis:** $\frac{1}{\sqrt{\text{ArcCosh}[z]}} = \frac{\text{Sinh}[\text{ArcCosh}[z]]}{\sqrt{\text{ArcCosh}[z]}} \text{ArcCosh}'[z]$
- **Rule:**

$$\int \frac{1}{\sqrt{\text{ArcCosh}[a+b\,x]}}\,dx \longrightarrow -\frac{\sqrt{\pi}\,\text{Erf}\left[\sqrt{\text{ArcCosh}[a+b\,x]}\right]}{2\,b} + \frac{\sqrt{\pi}\,\text{Erfi}\left[\sqrt{\text{ArcCosh}[a+b\,x]}\right]}{2\,b}$$

- **Program code:**

```
Int[1/Sqrt[ArcCosh[a_.+b_.*x_]],x_Symbol] :=
  -Sqrt[Pi]*Erf[Sqrt[ArcCosh[a+b*x]]]/(2*b) +
  Sqrt[Pi]*Erfi[Sqrt[ArcCosh[a+b*x]]]/(2*b) /;
FreeQ[{a,b},x]
```

- **Derivation: Integration by parts**
- **Rule:**

$$\int \sqrt{\text{ArcCosh}[a+b\,x]}\,dx \longrightarrow \frac{(a+b\,x)\sqrt{\text{ArcCosh}[a+b\,x]}}{b} - \frac{\sqrt{\pi}\,\text{Erf}\left[\sqrt{\text{ArcCosh}[a+b\,x]}\right]}{4\,b} - \frac{\sqrt{\pi}\,\text{Erfi}\left[\sqrt{\text{ArcCosh}[a+b\,x]}\right]}{4\,b}$$

- **Program code:**

```
Int[Sqrt[ArcCosh[a_.+b_.*x_]],x_Symbol] :=
  (a+b*x)*Sqrt[ArcCosh[a+b*x]]/b -
  Sqrt[Pi]*Erf[Sqrt[ArcCosh[a+b*x]]]/(4*b) -
  Sqrt[Pi]*Erfi[Sqrt[ArcCosh[a+b*x]]]/(4*b) /;
FreeQ[{a,b},x]
```

- **Derivation: Inverted iterated integration by parts**
- **Rule:** If $n < -1 \land n \neq -2$, then

$$\int \text{ArcCosh}[a+b\,x]^n\,dx \longrightarrow -\frac{(a+b\,x)\,\text{ArcCosh}[a+b\,x]^{n+2}}{b\,(n+1)\,(n+2)} + \frac{\sqrt{-1+a+b\,x}\,\sqrt{1+a+b\,x}\,\text{ArcCosh}[a+b\,x]^{n+1}}{b\,(n+1)} + \frac{1}{(n+1)\,(n+2)}\int \text{ArcCosh}[a+b\,x]^{n+2}\,dx$$

- **Program code:**

```
Int[ArcCosh[a_.+b_.*x_]^n_,x_Symbol] :=
  -(a+b*x)*ArcCosh[a+b*x]^(n+2)/(b*(n+1)*(n+2)) +
  Sqrt[-1+a+b*x]*Sqrt[1+a+b*x]*ArcCosh[a+b*x]^(n+1)/(b*(n+1)) +
  Dist[1/((n+1)*(n+2)),Int[ArcCosh[a+b*x]^(n+2),x]] /;
FreeQ[{a,b},x] && RationalQ[n] && n<-1 && n!=-2
```

- **Rule: If** $n \notin \mathbb{Q} \lor -1 < n < 1$**, then**

$$\int \text{ArcCosh}[a+b\,x]^n\,dx \longrightarrow \frac{\text{ArcCosh}[a+b\,x]^n\,\text{Gamma}[n+1,\,-\text{ArcCosh}[a+b\,x]]}{2\,b\,(-\text{ArcCosh}[a+b\,x])^n}+\frac{\text{Gamma}[n+1,\,\text{ArcCosh}[a+b\,x]]}{2\,b}$$

- **Program code:**

```
Int[ArcCosh[a_.+b_.*x_]^n_,x_Symbol] :=
  ArcCosh[a+b*x]^n*Gamma[n+1,-ArcCosh[a+b*x]]/(2*b*(-ArcCosh[a+b*x])^n) +
  Gamma[n+1,ArcCosh[a+b*x]]/(2*b) /;
FreeQ[{a,b,n},x] && (Not[RationalQ[n]] || -1<n<1)
```

$$\int x^m \operatorname{ArcCosh}[a + b x] \, dx$$

- **Reference: CRC 584, A&S 4.6.52**
- **Derivation: Integration by parts**
- **Rule: If $m + 1 \neq 0$, then**

$$\int x^m \operatorname{ArcCosh}[a + b x] \, dx \longrightarrow \frac{x^{m+1} \operatorname{ArcCosh}[a + b x]}{m + 1} - \frac{b}{m + 1} \int \frac{x^{m+1}}{\sqrt{-1 + a + b x} \sqrt{1 + a + b x}} \, dx$$

- **Program code:**

```
Int[x_^m_.*ArcCosh[a_.+b_.*x_],x_Symbol] :=
  x^(m+1)*ArcCosh[a+b*x]/(m+1) -
  Dist[b/(m+1),Int[x^(m+1)/(Sqrt[-1+a+b*x]*Sqrt[1+a+b*x]),x]] /;
FreeQ[{a,b,m},x] && NonzeroQ[m+1]
```

$$\int x^m \operatorname{ArcCosh}[a\,x]^n \, dx$$

- **Rule:**

$$\int \frac{x}{\sqrt{\operatorname{ArcCosh}[a\,x]}} \, dx \longrightarrow -\frac{1}{4\,a^2}\sqrt{\frac{\pi}{2}}\operatorname{Erf}\left[\sqrt{2}\,\sqrt{\operatorname{ArcCosh}[a\,x]}\right]+\frac{1}{4\,a^2}\sqrt{\frac{\pi}{2}}\operatorname{Erfi}\left[\sqrt{2}\,\sqrt{\operatorname{ArcCosh}[a\,x]}\right]$$

- **Program code:**

```
Int[x_/Sqrt[ArcCosh[a_.*x_]],x_Symbol] :=
  -Sqrt[Pi/2]*Erf[Sqrt[2]*Sqrt[ArcCosh[a*x]]]/(4*a^2) +
  Sqrt[Pi/2]*Erfi[Sqrt[2]*Sqrt[ArcCosh[a*x]]]/(4*a^2) /;
FreeQ[a,x]
```

- **Rule:**

$$\int \frac{x}{\operatorname{ArcCosh}[a\,x]^{3/2}} \, dx \longrightarrow$$
$$-\frac{2\,x\,\sqrt{-1+a\,x}\,\sqrt{1+a\,x}}{a\,\sqrt{\operatorname{ArcCosh}[a\,x]}}+\frac{1}{a^2}\sqrt{\frac{\pi}{2}}\operatorname{Erf}\left[\sqrt{2}\,\sqrt{\operatorname{ArcCosh}[a\,x]}\right]+\frac{1}{a^2}\sqrt{\frac{\pi}{2}}\operatorname{Erfi}\left[\sqrt{2}\,\sqrt{\operatorname{ArcCosh}[a\,x]}\right]$$

- **Program code:**

```
Int[x_/ArcCosh[a_.*x_]^(3/2),x_Symbol] :=
  -2*x*Sqrt[-1+a*x]*Sqrt[1+a*x]/(a*Sqrt[ArcCosh[a*x]]) +
   Sqrt[Pi/2]*Erf[Sqrt[2]*Sqrt[ArcCosh[a*x]]]/a^2 +
  Sqrt[Pi/2]*Erfi[Sqrt[2]*Sqrt[ArcCosh[a*x]]]/a^2 /;
FreeQ[a,x]
```

- **Rule:** If $n>1$, then

$$\int x \operatorname{ArcCosh}[a\,x]^n \, dx \longrightarrow -\frac{n\,x\,\sqrt{-1+a\,x}\,\sqrt{1+a\,x}\,\operatorname{ArcCosh}[a\,x]^{n-1}}{4\,a}-$$
$$\frac{\operatorname{ArcCosh}[a\,x]^n}{4\,a^2}+\frac{x^2\operatorname{ArcCosh}[a\,x]^n}{2}+\frac{n\,(n-1)}{4}\int x\operatorname{ArcCosh}[a\,x]^{n-2}\,dx$$

- **Program code:**

```
Int[x_*ArcCosh[a_.*x_]^n_,x_Symbol] :=
  -n*x*Sqrt[-1+a*x]*Sqrt[1+a*x]*ArcCosh[a*x]^(n-1)/(4*a) -
  ArcCosh[a*x]^n/(4*a^2) + x^2*ArcCosh[a*x]^n/2 +
  Dist[n*(n-1)/4,Int[x*ArcCosh[a*x]^(n-2),x]] /;
FreeQ[a,x] && RationalQ[n] && n>0
```

- **Rule: If $n < -1 \wedge n \neq -2$, then**

$$\int x\,\text{ArcCosh}[a\,x]^n\,dx \longrightarrow \frac{x\sqrt{-1+a\,x}\,\sqrt{1+a\,x}\,\text{ArcCosh}[a\,x]^{n+1}}{a\,(n+1)} + \frac{\text{ArcCosh}[a\,x]^{n+2}}{a^2\,(n+1)\,(n+2)} - \frac{2\,x^2\,\text{ArcCosh}[a\,x]^{n+2}}{(n+1)\,(n+2)} + \frac{4}{(n+1)\,(n+2)}\int x\,\text{ArcCosh}[a\,x]^{n+2}\,dx$$

- **Program code:**

```
Int[x_*ArcCosh[a_.*x_]^n_,x_Symbol] :=
  x*Sqrt[-1+a*x]*Sqrt[1+a*x]*ArcCosh[a*x]^(n+1)/(a*(n+1)) +
  ArcCosh[a*x]^(n+2)/(a^2*(n+1)*(n+2)) -
  2*x^2*ArcCosh[a*x]^(n+2)/((n+1)*(n+2)) +
  Dist[4/((n+1)*(n+2)),Int[x*ArcCosh[a*x]^(n+2),x]] /;
FreeQ[a,x] && RationalQ[n] && n<-1 && n≠-2
```

- **Rule: If $n > 1$, then**

$$\int \frac{\text{ArcCosh}[a\,x]^n}{x^3}\,dx \longrightarrow \frac{a\,n\sqrt{-1+a\,x}\,\sqrt{1+a\,x}\,\text{ArcCosh}[a\,x]^{n-1}}{2\,x} - \frac{\text{ArcCosh}[a\,x]^n}{2\,x^2} - \frac{a^2\,n\,(n-1)}{2}\int \frac{\text{ArcCosh}[a\,x]^{n-2}}{x}\,dx$$

- **Program code:**

```
Int[ArcCosh[a_.*x_]^n_/x_^3,x_Symbol] :=
  a*n*Sqrt[-1+a*x]*Sqrt[1+a*x]*ArcCosh[a*x]^(n-1)/(2*x) -
  ArcCosh[a*x]^n/(2*x^2) -
  Dist[a^2*n*(n-1)/2,Int[ArcCosh[a*x]^(n-2)/x,x]] /;
FreeQ[a,x] && RationalQ[n] && n>1
```

- **Rule: If** $m \in \mathbb{Z} \wedge m < -3 \wedge n > 1$**, then**

$$\int x^m \operatorname{ArcCosh}[a x]^n \, dx \longrightarrow \frac{a n x^{m+2} \sqrt{-1+a x} \sqrt{1+a x} \operatorname{ArcCosh}[a x]^{n-1}}{(m+1)(m+2)} + \frac{x^{m+1} \operatorname{ArcCosh}[a x]^n}{(m+1)} - \frac{a^2 (m+3) x^{m+3} \operatorname{ArcCosh}[a x]^n}{(m+1)(m+2)} + \frac{a^2 (m+3)^2}{(m+1)(m+2)} \int x^{m+2} \operatorname{ArcCosh}[a x]^n \, dx - \frac{a^2 n (n-1)}{(m+1)(m+2)} \int x^{m+2} \operatorname{ArcCosh}[a x]^{n-2} \, dx$$

- **Program code:**

```
Int[x_^m_*ArcCosh[a_.*x_]^n_,x_Symbol] :=
  a*n*x^(m+2)*Sqrt[-1+a*x]*Sqrt[1+a*x]*ArcCosh[a*x]^(n-1)/((m+1)*(m+2)) +
  x^(m+1)*ArcCosh[a*x]^n/(m+1) -
  a^2*(m+3)*x^(m+3)*ArcCosh[a*x]^n/((m+1)*(m+2)) +
  Dist[a^2*(m+3)^2/((m+1)*(m+2)),Int[x^(m+2)*ArcCosh[a*x]^n,x]] -
  Dist[a^2*n*(n-1)/((m+1)*(m+2)),Int[x^(m+2)*ArcCosh[a*x]^(n-2),x]] /;
FreeQ[a,x] && IntegerQ[m] && RationalQ[n] && m<-3 && n>1
```

- **Rule: If** $m \in \mathbb{Z} \wedge m > 1 \wedge n < -1 \wedge n \neq -2$**, then**

$$\int x^m \operatorname{ArcCosh}[a x]^n \, dx \longrightarrow \frac{x^m \sqrt{-1+a x} \sqrt{1+a x} \operatorname{ArcCosh}[a x]^{n+1}}{a (n+1)} + \frac{m x^{m-1} \operatorname{ArcCosh}[a x]^{n+2}}{a^2 (n+1)(n+2)} - \frac{(m+1) x^{m+1} \operatorname{ArcCosh}[a x]^{n+2}}{(n+1)(n+2)} + \frac{(m+1)^2}{(n+1)(n+2)} \int x^m \operatorname{ArcCosh}[a x]^{n+2} \, dx - \frac{m (m-1)}{a^2 (n+1)(n+2)} \int x^{m-2} \operatorname{ArcCosh}[a x]^{n+2} \, dx$$

- **Program code:**

```
Int[x_^m_*ArcCosh[a_.*x_]^n_,x_Symbol] :=
  x^m*Sqrt[-1+a*x]*Sqrt[1+a*x]*ArcCosh[a*x]^(n+1)/(a*(n+1)) +
  m*x^(m-1)*ArcCosh[a*x]^(n+2)/(a^2*(n+1)*(n+2)) -
  (m+1)*x^(m+1)*ArcCosh[a*x]^(n+2)/((n+1)*(n+2)) +
  Dist[(m+1)^2/((n+1)*(n+2)),Int[x^m*ArcCosh[a*x]^(n+2),x]] -
  Dist[m*(m-1)/(a^2*(n+1)*(n+2)),Int[x^(m-2)*ArcCosh[a*x]^(n+2),x]] /;
FreeQ[a,x] && IntegerQ[m] && RationalQ[n] && m>1 && n<-1 && n≠-2
```

- **Derivation: Integration by substitution**

- **Basis:** $\frac{\text{ArcCosh}[a\,x^p]^n}{x} = \frac{1}{p}\,\text{ArcCosh}[a\,x^p]^n\,\text{Tanh}[\text{ArcCosh}[a\,x^p]]\,\partial_x\text{ArcCosh}[a\,x^p]$

- **Rule: If** $n \in \mathbb{Z} \wedge n > 0$**, then**

$$\int \frac{\text{ArcCosh}[a\,x^p]^n}{x}\,dx \longrightarrow \frac{1}{p}\,\text{Subst}\left[\int x^n\,\text{Tanh}[x]\,dx,\ x,\ \text{ArcCosh}[a\,x^p]\right]$$

- **Program code:**

```
Int[ArcCosh[a_.*x_^p_.]^n_./x_,x_Symbol] :=
  Dist[1/p,Subst[Int[x^n*Tanh[x],x],x,ArcCosh[a*x^p]]] /;
FreeQ[{a,p},x] && IntegerQ[n] && n>0
```

- **Derivation: Integration by parts and substitution**

- **Basis: If** $m \in \mathbb{Z}$, $\frac{x^{m+1}\,\text{ArcCosh}[a\,x]^{n-1}}{\sqrt{-1+a\,x}\,\sqrt{1+a\,x}} = \frac{1}{a^{m+2}}\,\text{ArcCosh}[a\,x]^{n-1}\,\text{Cosh}[\text{ArcCosh}[a\,x]]^{m+1}\,\partial_x\text{ArcCosh}[a\,x]$

- **Rule: If** $m \in \mathbb{Z} \wedge m \neq -1$**, then**

$$\int x^m\,\text{ArcCosh}[a\,x]^n\,dx \longrightarrow \frac{x^{m+1}\,\text{ArcCosh}[a\,x]^n}{m+1} - \frac{n}{a^{m+1}\,(m+1)}\,\text{Subst}\left[\int x^{n-1}\,\text{Cosh}[x]^{m+1}\,dx,\ x,\ \text{ArcCosh}[a\,x]\right]$$

- **Program code:**

```
Int[x_^m_.*ArcCosh[a_.*x_]^n_,x_Symbol] :=
  x^(m+1)*ArcCosh[a*x]^n/(m+1) -
  Dist[n/(a^(m+1)*(m+1)),Subst[Int[x^(n-1)*Cosh[x]^(m+1),x],x,ArcCosh[a*x]]] /;
FreeQ[{a,n},x] && IntegerQ[m] && m≠-1
```

$$\int (a + b \,\text{ArcCosh}[c + d\,x])^n \, dx$$

- **Derivation: Integration by substitution**
- **Basis:** $(a + b\,\text{ArcCosh}[c + d\,x])^n = \frac{1}{d}\,(a + b\,\text{ArcCosh}[c + d\,x])^n\,\text{Sinh}[\text{ArcCosh}[c + d\,x]]\,\partial_x \text{ArcCosh}[c + d\,x]$
- **Rule: If** $n \notin \mathbb{Z}$**, then**

$$\int (a + b\,\text{ArcCosh}[c + d\,x])^n \, dx \longrightarrow \frac{1}{d}\,\text{Subst}\left[\int (a + b\,x)^n \,\text{Sinh}[x] \, dx,\ x,\ \text{ArcCosh}[c + d\,x]\right]$$

- **Program code:**

```
Int[(a_+b_.*ArcCosh[c_.+d_.*x_])^n_,x_Symbol] :=
  Dist[1/d,Subst[Int[(a+b*x)^n*Sinh[x],x],x,ArcCosh[c+d*x]]] /;
FreeQ[{a,b,c,d},x] && Not[IntegerQ[n]]
```

$$\int x^m (a + b\,\mathrm{ArcCosh}[c + d\,x])^n\,dx$$

- **Derivation: Integration by substitution**

- **Basis: If** $m \in \mathbb{Z}$, $x^m (a + b\,\mathrm{ArcCosh}[c + d\,x])^n =$
$$\frac{1}{d^{m+1}} (a + b\,\mathrm{ArcCosh}[c + d\,x])^n (\mathrm{Cosh}[\mathrm{ArcCosh}[c + d\,x]] - c)^m\,\mathrm{Sinh}[\mathrm{ArcCosh}[c + d\,x]]\,\partial_x \mathrm{ArcCosh}[c + d\,x]$$

- **Rule: If** $m \in \mathbb{Z} \bigwedge n \notin \mathbb{Z} \bigwedge m > 0$, **then**

$$\int x^m (a + b\,\mathrm{ArcCosh}[c + d\,x])^n\,dx \longrightarrow \frac{1}{d^{m+1}}\,\mathrm{Subst}\left[\int (a + b\,x)^n (\mathrm{Cosh}[x] - c)^m\,\mathrm{Sinh}[x]\,dx,\ x,\ \mathrm{ArcCosh}[c + d\,x]\right]$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*ArcCosh[c_.+d_.*x_])^n_,x_Symbol] :=
  Dist[1/d^(m+1),Subst[Int[(a+b*x)^n*(Cosh[x]-c)^m*Sinh[x],x],x,ArcCosh[c+d*x]]] /;
FreeQ[{a,b,c,d},x] && IntegerQ[m] && Not[IntegerQ[n]] && m>0
```

$$\int u\,\text{ArcCosh}\left[\frac{c}{a+b\,x^n}\right]^m dx$$

- **Derivation: Algebraic simplification**
- **Basis:** $\text{ArcCosh}[z] = \text{ArcSech}\left[\frac{1}{z}\right]$
- **Rule:**

$$\int u\,\text{ArcCosh}\left[\frac{c}{a+b\,x^n}\right]^m dx \longrightarrow \int u\,\text{ArcSech}\left[\frac{a}{c}+\frac{b\,x^n}{c}\right]^m dx$$

- **Program code:**

```
Int[u_.*ArcCosh[c_./(a_.+b_.*x_^n_.)]^m_.,x_Symbol] :=
  Int[u*ArcSech[a/c+b*x^n/c]^m,x] /;
FreeQ[{a,b,c,n,m},x]
```

$$\int f^{c\,\mathrm{ArcCosh}[a+b\,x]}\,dx$$

- **Rule: If** $1-c^2\,\mathrm{Log}[f]^2\neq 0$**, then**

$$\int f^{c\,\mathrm{ArcCosh}[a+b\,x]}\,dx\;\longrightarrow\;\frac{a+b\,x-c\,\sqrt{\frac{-1+a+b\,x}{1+a+b\,x}}\,(1+a+b\,x)\,\mathrm{Log}[f]}{b\,\left(1-c^2\,\mathrm{Log}[f]^2\right)}\,f^{c\,\mathrm{ArcCosh}[a+b\,x]}$$

- **Program code:**

```
Int[f_^(c_.*ArcCosh[a_.+b_.*x_]),x_Symbol] :=
  (a+b*x-c*Sqrt[(-1+a+b*x)/(1+a+b*x)]*(1+a+b*x)*Log[f])/(b*(1-c^2*Log[f]^2))*
    f^(c*ArcCosh[a+b*x]) /;
FreeQ[{a,b,c,f},x] && NonzeroQ[1-c^2*Log[f]^2]
```

$$\int \text{ArcCosh}[u] \, dx$$

- **Derivation: Integration by parts**

- **Rule: If u is free of inverse functions, then**

$$\int \text{ArcCosh}[u] \, dx \longrightarrow x \, \text{ArcCosh}[u] - \int \frac{x \, \partial_x u}{\sqrt{-1+u} \, \sqrt{1+u}} \, dx$$

- **Program code:**

```
Int[ArcCosh[u_],x_Symbol] :=
  x*ArcCosh[u] -
  Int[Regularize[x*D[u,x]/(Sqrt[-1+u]*Sqrt[1+u]),x],x] /;
InverseFunctionFreeQ[u,x] && Not[FunctionOfExponentialOfLinear[u,x]]
```

$$\int x^m\, e^{n\,\text{ArcCosh}[u]}\,dx$$

- **Derivation: Algebraic simplification**

- **Basis:** $e^{n\,\text{ArcCosh}[z]} = \left(z+\sqrt{-1+z}\,\sqrt{1+z}\right)^n$

- **Basis: If** $n\in\mathbb{Z}$, $e^{n\,\text{ArcCosh}[z]} == \left(z+\sqrt{\frac{-1+z}{1+z}}+z\sqrt{\frac{-1+z}{1+z}}\right)^n$

- **Rule: If** $n\in\mathbb{Z}\land u$ is a polynomial in x, **then**

$$\int e^{n\,\text{ArcCosh}[u]}\,dx \longrightarrow \int\left(u+\sqrt{-1+u}\,\sqrt{1+u}\right)^n dx$$

- **Program code:**

```
Int[E^(n_.*ArcCosh[u_]), x_Symbol] :=
  Int[(u+Sqrt[-1+u]*Sqrt[1+u])^n,x] /;
IntegerQ[n] && PolynomialQ[u,x]
```

- **Derivation: Algebraic simplification**

- **Basis:** $e^{n\,\text{ArcCosh}[z]} = \left(z+\sqrt{-1+z}\,\sqrt{1+z}\right)^n$

- **Rule: If** $n\in\mathbb{Z}\land u$ is a polynomial in x, **then**

$$\int x^m\, e^{n\,\text{ArcCosh}[u]}\,dx \longrightarrow \int x^m\left(u+\sqrt{-1+u}\,\sqrt{1+u}\right)^n dx$$

- **Program code:**

```
Int[x_^m_.*E^(n_.*ArcCosh[u_]), x_Symbol] :=
  Int[x^m*(u+Sqrt[-1+u]*Sqrt[1+u])^n,x] /;
RationalQ[m] && IntegerQ[n] && PolynomialQ[u,x]
```