$$\int x^m\,(a+b\,\mathrm{Sinh}[c+d\,x])^n\,dx$$

- **Derivation: Algebraic simplification**
- **Basis: If** $a^2+b^2=0$, **then** $a+b\,\mathrm{Sinh}[z]=2\,a\,\mathrm{Cosh}\left[-\frac{\pi a}{4b}+\frac{z}{2}\right]^2$
- **Rule: If** $a^2+b^2=0\,\wedge\,m\in\mathbb{Q}\,\wedge\,n\in\mathbb{Z}\,\wedge\,n<0$, **then**

$$\int x^m\,(a+b\,\mathrm{Sinh}[c+d\,x])^n\,dx\,\longrightarrow\,(2\,a)^n\int x^m\,\mathrm{Cosh}\left[-\frac{\pi a}{4b}+\frac{c}{2}+\frac{d\,x}{2}\right]^{2n}dx$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*Sinh[c_.+d_.*x_])^n_,x_Symbol] :=
  Dist[(2*a)^n,Int[x^m*Cosh[-Pi*a/(4*b)+c/2+d*x/2]^(2*n),x]] /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2+b^2] && RationalQ[m] && IntegerQ[n] && n<0
```

- **Derivation: Algebraic simplification and piecewise constant extraction**
- **Basis: If** $a^2+b^2=0$, **then** $a+b\,\mathrm{Sinh}[z]=2\,a\,\mathrm{Cosh}\left[-\frac{\pi a}{4b}+\frac{z}{2}\right]^2$
- **Basis: If** $a^2+b^2=0$, **then** $\partial_z\frac{\sqrt{a+b\,\mathrm{Sinh}[z]}}{\mathrm{Cosh}\left[-\frac{\pi a}{4b}+\frac{z}{2}\right]}=0$
- **Rule: If** $a^2+b^2=0\,\bigwedge\,m\in\mathbb{Q}\,\bigwedge\,n-\frac{1}{2}\in\mathbb{Z}$, **then**

$$\int x^m\,(a+b\,\mathrm{Sinh}[c+d\,x])^n\,dx\,\longrightarrow\,\frac{(2\,a)^{n-\frac{1}{2}}\sqrt{a+b\,\mathrm{Sinh}[c+d\,x]}}{\mathrm{Cosh}\left[-\frac{\pi a}{4b}+\frac{c}{2}+\frac{d\,x}{2}\right]}\int x^m\,\mathrm{Cosh}\left[-\frac{\pi a}{4b}+\frac{c}{2}+\frac{d\,x}{2}\right]^{2n}dx$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*Sinh[c_.+d_.*x_])^n_,x_Symbol] :=
  Dist[(2*a)^(n-1/2)*Sqrt[a+b*Sinh[c+d*x]]/Cosh[-Pi*a/(4*b)+c/2+d*x/2],
    Int[x^m*Cosh[-Pi*a/(4*b)+c/2+d*x/2]^(2*n),x]] /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2+b^2] && RationalQ[m] && IntegerQ[n-1/2]
```

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{1}{(a+b z)^2} = \frac{a}{\left(a^2+b^2\right) (a+b z)} + \frac{b (b-a z)}{\left(a^2+b^2\right) (a+b z)^2}$
- **Rule: If** $a^2 + b^2 \neq 0$**, then**

$$\int \frac{x}{(a + b \operatorname{Sinh}[c + d x])^2} \, dx \longrightarrow \frac{a}{a^2 + b^2} \int \frac{x}{a + b \operatorname{Sinh}[c + d x]} \, dx + \frac{b}{a^2 + b^2} \int \frac{x (b - a \operatorname{Sinh}[c + d x])}{(a + b \operatorname{Sinh}[c + d x])^2} \, dx$$

- **Program code:**

```
Int[x_/(a_+b_.*Sinh[c_.+d_.*x_])^2,x_Symbol] :=
  Dist[a/(a^2+b^2),Int[x/(a+b*Sinh[c+d*x]),x]] +
  Dist[b/(a^2+b^2),Int[x*(b-a*Sinh[c+d*x])/(a+b*Sinh[c+d*x])^2,x]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2+b^2]
```

- **Derivation: Algebraic expansion**
- **Basis:** $a + b \operatorname{Sinh}[z] = \frac{-b+2 a e^z+b e^{2 z}}{2 e^z}$
- **Rule: If** $a^2 + b^2 \neq 0 \bigwedge m > 0 \bigwedge n \in \mathbb{Z} \bigwedge n < 0$**, then**

$$\int x^m (a + b \operatorname{Sinh}[c + d x])^n \, dx \longrightarrow \frac{1}{2^n} \int \frac{x^m \left(-b + 2 a e^{c+d x} + b e^{2 (c+d x)}\right)^n}{e^{n (c+d x)}} \, dx$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*Sinh[c_.+d_.*x_])^n_,x_Symbol] :=
  Dist[1/2^n,Int[x^m*(-b+2*a*E^(c+d*x)+b*E^(2*(c+d*x)))^n/E^(n*(c+d*x)),x]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2+b^2] && RationalQ[m] && m>0 && IntegerQ[n] && n<0
```

$$\int x^m\,(a+b\,\mathrm{Cosh}[c+d\,x])^n\,dx$$

- **Derivation: Algebraic simplification**
- **Basis:** $1+\mathrm{Cosh}[z]=2\,\mathrm{Cosh}\left[\frac{z}{2}\right]^2$
- **Rule: If** $a-b=0\,\land\,m\in\mathbb{Q}\,\land\,n\in\mathbb{Z}\,\land\,n<0$**, then**

$$\int x^m\,(a+b\,\mathrm{Cosh}[c+d\,x])^n\,dx\,\longrightarrow\,(2\,a)^n\int x^m\,\mathrm{Cosh}\left[\frac{c}{2}+\frac{d\,x}{2}\right]^{2\,n}dx$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*Cosh[c_.+d_.*x_])^n_,x_Symbol] :=
  Dist[(2*a)^n,Int[x^m*Cosh[c/2+d*x/2]^(2*n),x]] /;
FreeQ[{a,b,c,d},x] && ZeroQ[a-b] && RationalQ[m] && IntegerQ[n] && n<0
```

- **Derivation: Algebraic simplification**
- **Basis:** $1-\mathrm{Cosh}[z]=-2\,\mathrm{Sinh}\left[\frac{z}{2}\right]^2$
- **Rule: If** $a+b=0\,\land\,m\in\mathbb{Q}\,\land\,n\in\mathbb{Z}\,\land\,n<0$**, then**

$$\int x^m\,(a+b\,\mathrm{Cosh}[c+d\,x])^n\,dx\,\longrightarrow\,(-2\,a)^n\int x^m\,\mathrm{Sinh}\left[\frac{c}{2}+\frac{d\,x}{2}\right]^{2\,n}dx$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*Cosh[c_.+d_.*x_])^n_,x_Symbol] :=
  Dist[(-2*a)^n,Int[x^m*Sinh[c/2+d*x/2]^(2*n),x]] /;
FreeQ[{a,b,c,d},x] && ZeroQ[a+b] && RationalQ[m] && IntegerQ[n] && n<0
```

- **Derivation: Algebraic simplification**
- **Basis:** $1 + \text{Cosh}[z] = 2\,\text{Cosh}\left[\frac{z}{2}\right]^2$
- **Basis:** $\partial_z \frac{\sqrt{a+a\,\text{Cosh}[z]}}{\text{Cosh}\left[\frac{z}{2}\right]} = 0$
- **Rule: If** $a - b = 0 \bigwedge m \in \mathbb{Q} \bigwedge n - \frac{1}{2} \in \mathbb{Z}$**, then**

$$\int x^m\,(a + b\,\text{Cosh}[c + d\,x])^n\,dx \longrightarrow \frac{(2\,a)^{n-\frac{1}{2}}\,\sqrt{a + b\,\text{Cosh}[c + d\,x]}}{\text{Cosh}\left[\frac{c}{2} + \frac{d\,x}{2}\right]}\int x^m\,\text{Cosh}\left[\frac{c}{2} + \frac{d\,x}{2}\right]^{2\,n}\,dx$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*Cosh[c_.+d_.*x_])^n_,x_Symbol] :=
  Dist[(2*a)^(n-1/2)*Sqrt[a+b*Cosh[c+d*x]]/Cosh[c/2+d*x/2],Int[x^m*Cosh[c/2+d*x/2]^(2*n),x]] /;
FreeQ[{a,b,c,d},x] && ZeroQ[a-b] && RationalQ[m] && IntegerQ[n-1/2]
```

- **Derivation: Algebraic simplification**
- **Basis:** $1 - \text{Cosh}[z] = -2\,\text{Sinh}\left[\frac{z}{2}\right]^2$
- **Basis:** $\partial_z \frac{\sqrt{a-a\,\text{Cosh}[z]}}{\text{Sinh}\left[\frac{z}{2}\right]} = 0$
- **Rule: If** $a + b = 0 \bigwedge m \in \mathbb{Q} \bigwedge n - \frac{1}{2} \in \mathbb{Z}$**, then**

$$\int x^m\,(a + b\,\text{Cosh}[c + d\,x])^n\,dx \longrightarrow \frac{(-2\,a)^{n-\frac{1}{2}}\,\sqrt{a + b\,\text{Cosh}[c + d\,x]}}{\text{Sinh}\left[\frac{c}{2} + \frac{d\,x}{2}\right]}\int x^m\,\text{Sinh}\left[\frac{c}{2} + \frac{d\,x}{2}\right]^{2\,n}\,dx$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*Cosh[c_.+d_.*x_])^n_,x_Symbol] :=
  Dist[(-2*a)^(n-1/2)*Sqrt[a+b*Cosh[c+d*x]]/Sinh[c/2+d*x/2],Int[x^m*Sinh[c/2+d*x/2]^(2*n),x]] /;
FreeQ[{a,b,c,d},x] && ZeroQ[a+b] && RationalQ[m] && IntegerQ[n-1/2]
```

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{1}{(a+b\,z)^2} = \frac{a}{\left(a^2-b^2\right)\,(a+b\,z)} - \frac{b\,(b+a\,z)}{\left(a^2-b^2\right)\,(a+b\,z)^2}$
- **Rule: If** $a^2 - b^2 \neq 0$, **then**

$$\int \frac{x}{(a+b\,\text{Cosh}[c+d\,x])^2}\,dx \,\longrightarrow\, \frac{a}{a^2-b^2}\int \frac{x}{a+b\,\text{Cosh}[c+d\,x]}\,dx - \frac{b}{a^2-b^2}\int \frac{x\,(b+a\,\text{Cosh}[c+d\,x])}{(a+b\,\text{Cosh}[c+d\,x])^2}\,dx$$

- **Program code:**

```
Int[x_/(a_+b_.*Cosh[c_.+d_.*x_])^2,x_Symbol] :=
  Dist[a/(a^2-b^2),Int[x/(a+b*Cosh[c+d*x]),x]] -
  Dist[b/(a^2-b^2),Int[x*(b+a*Cosh[c+d*x])/(a+b*Cosh[c+d*x])^2,x]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2-b^2]
```

- **Derivation: Algebraic expansion**
- **Basis:** $a + b\,\text{Cosh}[z] = \frac{b+2\,a\,e^z+b\,e^{2\,z}}{2\,e^z}$
- **Rule: If** $a^2 - b^2 \neq 0 \wedge m > 0 \wedge n \in \mathbb{Z} \wedge n < 0$, **then**

$$\int x^m\,(a+b\,\text{Cosh}[c+d\,x])^n\,dx \,\longrightarrow\, \frac{1}{2^n}\int \frac{x^m\left(b+2\,a\,e^{c+d\,x}+b\,e^{2\,(c+d\,x)}\right)^n}{e^{n\,(c+d\,x)}}\,dx$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*Cosh[c_.+d_.*x_])^n_,x_Symbol] :=
  Dist[1/2^n,Int[x^m*(b+2*a*E^(c+d*x)+b*E^(2*(c+d*x)))^n/E^(n*(c+d*x)),x]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2-b^2] && RationalQ[m] && m>0 && IntegerQ[n] && n<0
```

$\int u\,(a+b\,\mathrm{Sinh}[c+d\,x]^2)^n\,dx$

- **Derivation: Algebraic simplification**
- **Basis:** $\mathrm{Sinh}[z]^2 = \frac{1}{2}\,(-1+\mathrm{Cosh}[2\,z])$
- **Note: This rule should be replaced with rules that directly reduce the integrand rather than transforming it using hyperbolic power expansion!**
- **Rule: If** $a-b\neq 0\,\land\,n\neq -1$**, then**

$$\int (a+b\,\mathrm{Sinh}[c+d\,x]^2)^n\,dx \;\longrightarrow\; \frac{1}{2^n}\int (2\,a-b+b\,\mathrm{Cosh}[2\,c+2\,d\,x])^n\,dx$$

- **Program code:**

```
Int[(a_+b_.*Sinh[c_.+d_.*x_]^2)^n_,x_Symbol] :=
  Dist[1/2^n,Int[(2*a-b+b*Cosh[2*c+2*d*x])^n,x]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a-b] && RationalQ[n] && n≠-1
```

- **Basis:** $\mathrm{Cosh}[z]^2 = \frac{1}{2}\,(1+\mathrm{Cosh}[2\,z])$

```
Int[(a_+b_.*Cosh[c_.+d_.*x_]^2)^n_,x_Symbol] :=
  Dist[1/2^n,Int[(2*a+b+b*Cosh[2*c+2*d*x])^n,x]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a+b] && RationalQ[n] && n≠-1
```

- **Derivation: Algebraic simplification**
- **Basis:** $\mathrm{Sinh}[z]^2 = \frac{1}{2}\,(-1+\mathrm{Cosh}[2\,z])$
- **Note: This rule should be replaced with rules that directly reduce the integrand rather than transforming it using hyperbolic power expansion!**
- **Rule: If** $a-b\neq 0\,\land\,m,\,n\in\mathbb{Z}\,\land\,m>0$**, then**

$$\int x^m\,(a+b\,\mathrm{Sinh}[c+d\,x]^2)^n\,dx \;\longrightarrow\; \frac{1}{2^n}\int x^m\,(2\,a-b+b\,\mathrm{Cosh}[2\,c+2\,d\,x])^n\,dx$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*Sinh[c_.+d_.*x_]^2)^n_,x_Symbol] :=
  Dist[1/2^n,Int[x^m*(2*a-b+b*Cosh[2*c+2*d*x])^n,x]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a-b] && IntegersQ[m,n] && (m>0 && n==-1 || m==1 && n==-2)
```

- **Basis:** $\mathrm{Cosh}[z]^2 = \frac{1}{2}\,(1+\mathrm{Cosh}[2\,z])$

```
Int[x_^m_.*(a_+b_.*Cosh[c_.+d_.*x_]^2)^n_,x_Symbol] :=
  Dist[1/2^n,Int[x^m*(2*a+b+b*Cosh[2*c+2*d*x])^n,x]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a+b] && IntegersQ[m,n] && (m>0 && n==-1 || m==1 && n==-2)
```

$$\int \text{Sinh}[a + b\,x^n]\, dx$$

- **Derivation: Primitive rule**
- **Basis:** $\text{FresnelS}'[z] = -i\,\text{Sinh}\left[\frac{i\,\pi\,z^2}{2}\right]$
- **Note: This rule is commented out since it introduces the imaginary unit i; whereas, converting the hyperbolic sine to exponential form does not.**
- **Rule:**

$$\int \text{Sinh}\left[b\,x^2\right] dx \longrightarrow -\frac{i\sqrt{\frac{\pi}{2}}}{\sqrt{i\,b}}\,\text{FresnelS}\left[\frac{\sqrt{i\,b}\;x}{\sqrt{\frac{\pi}{2}}}\right]$$

- **Program code:**

```
(* Int[Sinh[b_.*x_^2],x_Symbol] :=
  -I*Sqrt[Pi/2]*FresnelS[Rt[I*b,2]*x/Sqrt[Pi/2]]/Rt[I*b,2] /;
FreeQ[b,x] *)
```

- **Basis:** $\text{FresnelC}'[z] = \text{Cosh}\left[\frac{I\,\pi\,z^2}{2}\right]$

```
(* Int[Cosh[b_.*x_^2],x_Symbol] :=
  Sqrt[Pi/2]*FresnelC[Rt[I*b,2]*x/Sqrt[Pi/2]]/Rt[I*b,2] /;
FreeQ[b,x] *)
```

- **Derivation: Algebraic expansion**
- **Basis:** $\text{Sinh}[z] = \frac{e^z}{2} - \frac{e^{-z}}{2}$
- **Rule: If** $\neg\,(n \in \mathbb{F} \lor n < 0)$**, then**

$$\int \text{Sinh}[a + b\,x^n]\, dx \longrightarrow \frac{1}{2}\int e^{a+b\,x^n}\, dx - \frac{1}{2}\int e^{-a-b\,x^n}\, dx$$

- **Program code:**

```
Int[Sinh[a_.+b_.*x_^n_],x_Symbol] :=
  Dist[1/2,Int[E^(a+b*x^n),x]] -
  Dist[1/2,Int[E^(-a-b*x^n),x]] /;
FreeQ[{a,b,n},x] && Not[FractionOrNegativeQ[n]]
```

```
Int[Cosh[a_.+b_.*x_^n_],x_Symbol] :=
  Dist[1/2,Int[E^(-a-b*x^n),x]] +
  Dist[1/2,Int[E^(a+b*x^n),x]] /;
FreeQ[{a,b,n},x] && Not[FractionOrNegativeQ[n]]
```

- **Derivation: Integration by parts**
- **Note: Although resulting integrand looks more complicated than the original, rules for improper binomials rectify it.**
- **Rule: If $n \in \mathbb{Z} \lor n < 0$, then**

$$\int \text{Sinh}[a + b\,x^n]\,dx \longrightarrow x\,\text{Sinh}[a + b\,x^n] - b\,n \int x^n\,\text{Cosh}[a + b\,x^n]\,dx$$

- **Program code:**

```
Int[Sinh[a_.+b_.*x_^n_],x_Symbol] :=
  x*Sinh[a+b*x^n] -
  Dist[b*n,Int[x^n*Cosh[a+b*x^n],x]] /;
FreeQ[{a,b},x] && IntegerQ[n] && n<0
```

```
Int[Cosh[a_.+b_.*x_^n_],x_Symbol] :=
  x*Cosh[a+b*x^n] -
  Dist[b*n,Int[x^n*Sinh[a+b*x^n],x]] /;
FreeQ[{a,b},x] && IntegerQ[n] && n<0
```

$$\int x^m \operatorname{Sinh}[a + b x^n] \, dx$$

- **Derivation: Primitive rule**
- **Basis:** $\operatorname{SinhIntegral}'[z] = \frac{\operatorname{Sinh}[z]}{z}$
- **Rule:**

$$\int \frac{\operatorname{Sinh}[b x^n]}{x} \, dx \longrightarrow \frac{\operatorname{SinhIntegral}[b x^n]}{n}$$

- **Program code:**

```
Int[Sinh[b_.*x_^n_.]/x_,x_Symbol] :=
  SinhIntegral[b*x^n]/n /;
FreeQ[{b,n},x]
```

- **Basis:** $\operatorname{CoshIntegral}'[z] = \frac{\operatorname{Cosh}[z]}{z}$

```
Int[Cosh[b_.*x_^n_.]/x_,x_Symbol] :=
  CoshIntegral[b*x^n]/n /;
FreeQ[{b,n},x]
```

- **Derivation: Algebraic expansion**
- **Basis:** $\operatorname{Sinh}[w + z] = \operatorname{Sinh}[w] \operatorname{Cosh}[z] + \operatorname{Cosh}[w] \operatorname{Sinh}[z]$
- **Rule:**

$$\int \frac{\operatorname{Sinh}[a + b x^n]}{x} \, dx \longrightarrow \operatorname{Sinh}[a] \int \frac{\operatorname{Cosh}[b x^n]}{x} \, dx + \operatorname{Cosh}[a] \int \frac{\operatorname{Sinh}[b x^n]}{x} \, dx$$

- **Program code:**

```
Int[Sinh[a_+b_.*x_^n_.]/x_,x_Symbol] :=
  Dist[Sinh[a],Int[Cosh[b*x^n]/x,x]] +
  Dist[Cosh[a],Int[Sinh[b*x^n]/x,x]] /;
FreeQ[{a,b,n},x]
```

```
Int[Cosh[a_+b_.*x_^n_.]/x_,x_Symbol] :=
  Dist[Cosh[a],Int[Cosh[b*x^n]/x,x]] +
  Dist[Sinh[a],Int[Sinh[b*x^n]/x,x]] /;
FreeQ[{a,b,n},x]
```

- **Reference: CRC 392h, A&S 4.5.83**
- **Derivation: Integration by parts**
- **Basis:** $x^m \operatorname{Sinh}[a+b\,x^n] = \frac{x^{m-n+1}\,\partial_x \operatorname{Cosh}[a+b\,x^n]}{b\,n}$
- **Rule: If** $n \in \mathbb{Z} \land 0 < n \leq m$**, then**

$$\int x^m \operatorname{Sinh}[a+b\,x^n]\,dx \longrightarrow \frac{x^{m-n+1}\operatorname{Cosh}[a+b\,x^n]}{b\,n} - \frac{m-n+1}{b\,n}\int x^{m-n}\operatorname{Cosh}[a+b\,x^n]\,dx$$

- **Program code:**

```
Int[x_^m_.*Sinh[a_.+b_.*x_^n_.],x_Symbol] :=
  x^(m-n+1)*Cosh[a+b*x^n]/(b*n) -
  Dist[(m-n+1)/(b*n),Int[x^(m-n)*Cosh[a+b*x^n],x]] /;
FreeQ[{a,b},x] && IntegerQ[n] && RationalQ[m] && 0<n<=m
```

- **Reference: CRC 396h, A&S 4.5.84**

```
Int[x_^m_.*Cosh[a_.+b_.*x_^n_.],x_Symbol] :=
  x^(m-n+1)*Sinh[a+b*x^n]/(b*n) -
  Dist[(m-n+1)/(b*n),Int[x^(m-n)*Sinh[a+b*x^n],x]] /;
FreeQ[{a,b},x] && IntegerQ[n] && RationalQ[m] && 0<n<=m
```

- **Reference: CRC 405h**
- **Derivation: Integration by parts**
- **Rule: If** $m+n+1=0 \lor (n \in \mathbb{Z} \land ((n>0 \land m<-1) \lor 0<-n<m+1))$**, then**

$$\int x^m \operatorname{Sinh}[a+b\,x^n]\,dx \longrightarrow \frac{x^{m+1}\operatorname{Sinh}[a+b\,x^n]}{m+1} - \frac{b\,n}{m+1}\int x^{m+n}\operatorname{Cosh}[a+b\,x^n]\,dx$$

- **Program code:**

```
Int[x_^m_.*Sinh[a_.+b_.*x_^n_.],x_Symbol] :=
  x^(m+1)*Sinh[a+b*x^n]/(m+1) -
  Dist[b*n/(m+1),Int[x^(m+n)*Cosh[a+b*x^n],x]] /;
FreeQ[{a,b,m,n},x] && (ZeroQ[m+n+1] || IntegerQ[n] && RationalQ[m] && (n>0 && m<-1 || 0<-n<m+1))
```

- **Reference: CRC 406h**

```
Int[x_^m_.*Cosh[a_.+b_.*x_^n_.],x_Symbol] :=
  x^(m+1)*Cosh[a+b*x^n]/(m+1) -
  Dist[b*n/(m+1),Int[x^(m+n)*Sinh[a+b*x^n],x]] /;
FreeQ[{a,b,m,n},x] && (ZeroQ[m+n+1] || IntegerQ[n] && RationalQ[m] && (n>0 && m<-1 || 0<-n<m+1))
```

- **Derivation: Algebraic expansion**
- **Basis:** $\text{Sinh}[z] = \frac{e^z}{2} - \frac{e^{-z}}{2}$
- **Rule: If** $m+1 \neq 0 \land m-n+1 \neq 0 \land \neg (m \in \mathbb{F} \lor n \in \mathbb{F} \lor n < 0)$**, then**

$$\int x^m \,\text{Sinh}[a + b\, x^n]\, dx \longrightarrow \frac{1}{2}\int x^m\, e^{a+b\, x^n}\, dx - \frac{1}{2}\int x^m\, e^{-a-b\, x^n}\, dx$$

- **Program code:**

```
Int[x_^m_.*Sinh[a_.+b_.*x_^n_.],x_Symbol] :=
  Dist[1/2,Int[x^m*E^(a+b*x^n),x]] -
  Dist[1/2,Int[x^m*E^(-a-b*x^n),x]] /;
FreeQ[{a,b,m,n},x] && NonzeroQ[m+1] && NonzeroQ[m-n+1] &&
Not[FractionQ[m] || FractionOrNegativeQ[n]]
```

```
Int[x_^m_.*Cosh[a_.+b_.*x_^n_.],x_Symbol] :=
  Dist[1/2,Int[x^m*E^(-a-b*x^n),x]] +
  Dist[1/2,Int[x^m*E^(a+b*x^n),x]] /;
FreeQ[{a,b,m,n},x] && NonzeroQ[m+1] && NonzeroQ[m-n+1] &&
Not[FractionQ[m] || FractionOrNegativeQ[n]]
```

$$\int x^m \operatorname{Sinh}[a + b\,x^n]^p\, dx$$

- **Derivation: Integration by parts**
- **Rule: If** $n, p \in \mathbb{Z} \land p > 1 \land n - 1 \neq 0$, **then**

$$\int \frac{\operatorname{Sinh}[a + b\,x^n]^p}{x^n}\, dx \longrightarrow -\frac{\operatorname{Sinh}[a + b\,x^n]^p}{(n-1)\,x^{n-1}} + \frac{b\,n\,p}{n-1} \int \operatorname{Sinh}[a + b\,x^n]^{p-1} \operatorname{Cosh}[a + b\,x^n]\, dx$$

- **Program code:**

```
Int[x_^m_.*Sinh[a_.+b_.*x_^n_.]^p_,x_Symbol] :=
  -Sinh[a+b*x^n]^p/((n-1)*x^(n-1)) +
  Dist[b*n*p/(n-1),Int[Sinh[a+b*x^n]^(p-1)*Cosh[a+b*x^n],x]] /;
FreeQ[{a,b},x] && IntegersQ[n,p] && ZeroQ[m+n] && p>1 && NonzeroQ[n-1]
```

```
Int[x_^m_.*Cosh[a_.+b_.*x_^n_.]^p_,x_Symbol] :=
  -Cosh[a+b*x^n]^p/((n-1)*x^(n-1)) +
  Dist[b*n*p/(n-1),Int[Cosh[a+b*x^n]^(p-1)*Sinh[a+b*x^n],x]] /;
FreeQ[{a,b},x] && IntegersQ[n,p] && ZeroQ[m+n] && p>1 && NonzeroQ[n-1]
```

- **Reference: G&R 2.471.1b' special case when** $m - 2\,n + 1 = 0$
- **Rule: If** $p > 1 \land m - 2\,n + 1 = 0$, **then**

$$\int x^m \operatorname{Sinh}[a + b\,x^n]^p\, dx \longrightarrow -\frac{n \operatorname{Sinh}[a + b\,x^n]^p}{b^2\,n^2\,p^2} + \frac{x^n \operatorname{Cosh}[a + b\,x^n] \operatorname{Sinh}[a + b\,x^n]^{p-1}}{b\,n\,p} - \frac{p-1}{p} \int x^m \operatorname{Sinh}[a + b\,x^n]^{p-2}\, dx$$

- **Program code:**

```
Int[x_^m_.*Sinh[a_.+b_.*x_^n_.]^p_,x_Symbol] :=
  -n*Sinh[a+b*x^n]^p/(b^2*n^2*p^2) +
  x^n*Cosh[a+b*x^n]*Sinh[a+b*x^n]^(p-1)/(b*n*p) -
  Dist[(p-1)/p,Int[x^m*Sinh[a+b*x^n]^(p-2),x]] /;
FreeQ[{a,b,m,n},x] && RationalQ[p] && p>1 && ZeroQ[m-2*n+1]
```

- **Reference: G&R 2.471.1a' special case with** $m - 2\,n + 1 = 0$

```
Int[x_^m_.*Cosh[a_.+b_.*x_^n_.]^p_,x_Symbol] :=
  -n*Cosh[a+b*x^n]^p/(b^2*n^2*p^2) +
  x^n*Sinh[a+b*x^n]*Cosh[a+b*x^n]^(p-1)/(b*n*p) +
  Dist[(p-1)/p,Int[x^m*Cosh[a+b*x^n]^(p-2),x]] /;
FreeQ[{a,b,m,n},x] && RationalQ[p] && p>1 && ZeroQ[m-2*n+1]
```

- **Reference: G&R 2.471.1b'**
- **Rule: If** $m, n \in \mathbb{Z} \land p > 1 \land 0 < 2n < m+1$, **then**

$$\int x^m \operatorname{Sinh}[a+b\,x^n]^p\,dx \longrightarrow -\frac{(m-n+1)\,x^{m-2n+1}\operatorname{Sinh}[a+b\,x^n]^p}{b^2\,n^2\,p^2}+\frac{x^{m-n+1}\operatorname{Cosh}[a+b\,x^n]\operatorname{Sinh}[a+b\,x^n]^{p-1}}{b\,n\,p}-\frac{p-1}{p}\int x^m\operatorname{Sinh}[a+b\,x^n]^{p-2}\,dx+\frac{(m-n+1)\,(m-2n+1)}{b^2\,n^2\,p^2}\int x^{m-2n}\operatorname{Sinh}[a+b\,x^n]^p\,dx$$

- **Program code:**

```
Int[x_^m_.*Sinh[a_.+b_.*x_^n_.]^p_,x_Symbol] :=
  -(m-n+1)*x^(m-2*n+1)*Sinh[a+b*x^n]^p/(b^2*n^2*p^2) +
  x^(m-n+1)*Cosh[a+b*x^n]*Sinh[a+b*x^n]^(p-1)/(b*n*p) -
  Dist[(p-1)/p,Int[x^m*Sinh[a+b*x^n]^(p-2),x]] +
  Dist[(m-n+1)*(m-2*n+1)/(b^2*n^2*p^2),Int[x^(m-2*n)*Sinh[a+b*x^n]^p,x]] /;
FreeQ[{a,b},x] && IntegersQ[m,n] && RationalQ[p] && p>1 && 0<2*n<m+1
```

- **Reference: G&R 2.631.3'**

```
Int[x_^m_.*Cosh[a_.+b_.*x_^n_.]^p_,x_Symbol] :=
  -(m-n+1)*x^(m-2*n+1)*Cosh[a+b*x^n]^p/(b^2*n^2*p^2) +
  x^(m-n+1)*Sinh[a+b*x^n]*Cosh[a+b*x^n]^(p-1)/(b*n*p) +
  Dist[(p-1)/p,Int[x^m*Cosh[a+b*x^n]^(p-2),x]] +
  Dist[(m-n+1)*(m-2*n+1)/(b^2*n^2*p^2),Int[x^(m-2*n)*Cosh[a+b*x^n]^p,x]] /;
FreeQ[{a,b},x] && IntegersQ[m,n] && RationalQ[p] && p>1 && 0<2*n<m+1
```

- **Reference: G&R 2.477.1 special case when** $m-2n+1=0$
- **Rule: If** $p < -1 \land p \neq -2 \land m-2n+1=0$, **then**

$$\int x^m \operatorname{Sinh}[a+b\,x^n]^p\,dx \longrightarrow \frac{x^n\operatorname{Cosh}[a+b\,x^n]\operatorname{Sinh}[a+b\,x^n]^{p+1}}{b\,n\,(p+1)}-\frac{n\operatorname{Sinh}[a+b\,x^n]^{p+2}}{b^2\,n^2\,(p+1)\,(p+2)}-\frac{p+2}{p+1}\int x^m\operatorname{Sinh}[a+b\,x^n]^{p+2}\,dx$$

- **Program code:**

```
Int[x_^m_.*Sinh[a_.+b_.*x_^n_.]^p_,x_Symbol] :=
  x^n*Cosh[a+b*x^n]*Sinh[a+b*x^n]^(p+1)/(b*n*(p+1)) -
  n*Sinh[a+b*x^n]^(p+2)/(b^2*n^2*(p+1)*(p+2)) -
  Dist[(p+2)/(p+1),Int[x^m*Sinh[a+b*x^n]^(p+2),x]] /;
FreeQ[{a,b,m,n},x] && RationalQ[p] && p<-1 && p≠-2 && ZeroQ[m-2*n+1]
```

- **Reference: G&R 2.477.2' special case with $m - 2n + 1 = 0$**

```
Int[x_^m_.*Cosh[a_.+b_.*x_^n_.]^p_,x_Symbol] :=
  -x^n*Sinh[a+b*x^n]*Cosh[a+b*x^n]^(p+1)/(b*n*(p+1)) +
  n*Cosh[a+b*x^n]^(p+2)/(b^2*n^2*(p+1)*(p+2)) +
  Dist[(p+2)/(p+1),Int[x^m*Cosh[a+b*x^n]^(p+2),x]] /;
FreeQ[{a,b,m,n},x] && RationalQ[p] && p<-1 && p≠-2 && ZeroQ[m-2*n+1]
```

- **Reference: G&R 2.477.1**

- **Rule: If $m, n \in \mathbb{Z} \wedge p < -1 \wedge p \neq -2 \wedge 0 < 2n < m + 1$, then**

$$\int x^m \operatorname{Sinh}[a + b x^n]^p \, dx \longrightarrow \frac{x^{m-n+1} \operatorname{Cosh}[a + b x^n] \operatorname{Sinh}[a + b x^n]^{p+1}}{b n (p + 1)} - \frac{(m - n + 1) x^{m-2n+1} \operatorname{Sinh}[a + b x^n]^{p+2}}{b^2 n^2 (p + 1) (p + 2)} - \frac{p + 2}{p + 1} \int x^m \operatorname{Sinh}[a + b x^n]^{p+2} \, dx + \frac{(m - n + 1) (m - 2n + 1)}{b^2 n^2 (p + 1) (p + 2)} \int x^{m-2n} \operatorname{Sinh}[a + b x^n]^{p+2} \, dx$$

- **Program code:**

```
Int[x_^m_.*Sinh[a_.+b_.*x_^n_.]^p_,x_Symbol] :=
  x^(m-n+1)*Cosh[a+b*x^n]*Sinh[a+b*x^n]^(p+1)/(b*n*(p+1)) -
  (m-n+1)*x^(m-2*n+1)*Sinh[a+b*x^n]^(p+2)/(b^2*n^2*(p+1)*(p+2)) -
  Dist[(p+2)/(p+1),Int[x^m*Sinh[a+b*x^n]^(p+2),x]] +
  Dist[(m-n+1)*(m-2*n+1)/(b^2*n^2*(p+1)*(p+2)),Int[x^(m-2*n)*Sinh[a+b*x^n]^(p+2),x]] /;
FreeQ[{a,b},x] && IntegersQ[m,n] && RationalQ[p] && p<-1 && p≠-2 && 0<2*n<m+1
```

- **Reference: G&R 2.477.2**

```
Int[x_^m_.*Cosh[a_.+b_.*x_^n_.]^p_,x_Symbol] :=
  -x^(m-n+1)*Sinh[a+b*x^n]*Cosh[a+b*x^n]^(p+1)/(b*n*(p+1)) +
  (m-n+1)*x^(m-2*n+1)*Cosh[a+b*x^n]^(p+2)/(b^2*n^2*(p+1)*(p+2)) +
  Dist[(p+2)/(p+1),Int[x^m*Cosh[a+b*x^n]^(p+2),x]] -
  Dist[(m-n+1)*(m-2*n+1)/(b^2*n^2*(p+1)*(p+2)),Int[x^(m-2*n)*Cosh[a+b*x^n]^(p+2),x]] /;
FreeQ[{a,b},x] && IntegersQ[m,n] && RationalQ[p] && p<-1 && p≠-2 && 0<2*n<m+1
```

- **Reference: G&R 2.475.1'**

- **Rule: If** $m, n \in \mathbb{Z} \land p > 1 \land 0 < 2n < 1 - m \land m + n + 1 \neq 0$**, then**

$$\int x^m \operatorname{Sinh}[a + b x^n]^p \, dx \longrightarrow \frac{x^{m+1} \operatorname{Sinh}[a + b x^n]^p}{m+1} - \frac{b n p x^{m+n+1} \operatorname{Cosh}[a + b x^n] \operatorname{Sinh}[a + b x^n]^{p-1}}{(m+1)(m+n+1)} + \frac{b^2 n^2 p^2}{(m+1)(m+n+1)} \int x^{m+2n} \operatorname{Sinh}[a + b x^n]^p \, dx + \frac{b^2 n^2 p (p-1)}{(m+1)(m+n+1)} \int x^{m+2n} \operatorname{Sinh}[a + b x^n]^{p-2} \, dx$$

- **Program code:**

```
Int[x_^m_.*Sinh[a_.+b_.*x_^n_.]^p_,x_Symbol] :=
  x^(m+1)*Sinh[a+b*x^n]^p/(m+1) -
  b*n*p*x^(m+n+1)*Cosh[a+b*x^n]*Sinh[a+b*x^n]^(p-1)/((m+1)*(m+n+1)) +
  Dist[b^2*n^2*p^2/((m+1)*(m+n+1)),Int[x^(m+2*n)*Sinh[a+b*x^n]^p,x]] +
  Dist[b^2*n^2*p*(p-1)/((m+1)*(m+n+1)),Int[x^(m+2*n)*Sinh[a+b*x^n]^(p-2),x]] /;
FreeQ[{a,b},x] && IntegersQ[m,n] && RationalQ[p] && p>1 && 0<2*n<-m+1 && NonzeroQ[m+n+1]
```

- **Reference: G&R 2.475.2'**

```
Int[x_^m_.*Cosh[a_.+b_.*x_^n_.]^p_,x_Symbol] :=
  x^(m+1)*Cosh[a+b*x^n]^p/(m+1) -
  b*n*p*x^(m+n+1)*Sinh[a+b*x^n]*Cosh[a+b*x^n]^(p-1)/((m+1)*(m+n+1)) +
  Dist[b^2*n^2*p^2/((m+1)*(m+n+1)),Int[x^(m+2*n)*Cosh[a+b*x^n]^p,x]] -
  Dist[b^2*n^2*p*(p-1)/((m+1)*(m+n+1)),Int[x^(m+2*n)*Cosh[a+b*x^n]^(p-2),x]] /;
FreeQ[{a,b},x] && IntegersQ[m,n] && RationalQ[p] && p>1 && 0<2*n<-m+1 && NonzeroQ[m+n+1]
```

$$\int x^m \operatorname{Sinh}\left[a+b\,(c+d\,x)^n\right]^p\,dx$$

- **Derivation: Integration by linear substitution**
- **Rule: If** $m \in \mathbb{Z} \land m > 0 \land p \in \mathbb{Q}$, **then**

$$\int x^m \operatorname{Sinh}\left[a+b\,(c+d\,x)^n\right]^p\,dx \longrightarrow \frac{1}{d}\,\text{Subst}\left[\int\left(-\frac{c}{d}+\frac{x}{d}\right)^m \operatorname{Sinh}\left[a+b\,x^n\right]^p\,dx,\;x,\;c+d\,x\right]$$

- **Program code:**

```
Int[x_^m_.*Sinh[a_.+b_.*(c_+d_.*x_)^n_]^p_.,x_Symbol] :=
  Dist[1/d,Subst[Int[(-c/d+x/d)^m*Sinh[a+b*x^n]^p,x],x,c+d*x]] /;
FreeQ[{a,b,c,d,n},x] && IntegerQ[m] && m>0 && RationalQ[p]
```

```
Int[x_^m_.*Cosh[a_.+b_.*(c_+d_.*x_)^n_]^p_.,x_Symbol] :=
  Dist[1/d,Subst[Int[(-c/d+x/d)^m*Cosh[a+b*x^n]^p,x],x,c+d*x]] /;
FreeQ[{a,b,c,d,n},x] && IntegerQ[m] && m>0 && RationalQ[p]
```

$$\int \text{Sinh}\left[a+b\,x+c\,x^2\right]\,dx$$

- **Derivation: Algebraic simplification**
- **Basis: If** $b^2-4\,a\,c=0$**, then** $a+b\,x+c\,x^2=\frac{(b+2\,c\,x)^2}{4\,c}$
- **Rule: If** $b^2-4\,a\,c=0$**, then**

$$\int \text{Sinh}\left[a+b\,x+c\,x^2\right]\,dx\;\longrightarrow\;\int \text{Sinh}\left[\frac{(b+2\,c\,x)^2}{4\,c}\right]\,dx$$

- **Program code:**

```
Int[Sinh[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  Int[Sinh[(b+2*c*x)^2/(4*c)],x] /;
FreeQ[{a,b,c},x] && ZeroQ[b^2-4*a*c]
```

```
Int[Cosh[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  Int[Cosh[(b+2*c*x)^2/(4*c)],x] /;
FreeQ[{a,b,c},x] && ZeroQ[b^2-4*a*c]
```

- **Derivation: Algebraic expansion**
- **Basis:** $\text{Sinh}[z]=\frac{e^z}{2}-\frac{e^{-z}}{2}$
- **Rule: If** $b^2-4\,a\,c\neq 0$**, then**

$$\int \text{Sinh}\left[a+b\,x+c\,x^2\right]\,dx\;\longrightarrow\;\frac{1}{2}\int e^{a+b\,x+c\,x^2}\,dx-\frac{1}{2}\int e^{-a-b\,x-c\,x^2}\,dx$$

- **Program code:**

```
Int[Sinh[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  Dist[1/2,Int[E^(a+b*x+c*x^2),x]] -
  Dist[1/2,Int[E^(-a-b*x-c*x^2),x]] /;
FreeQ[{a,b,c},x] && NonzeroQ[b^2-4*a*c]
```

```
Int[Cosh[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  Dist[1/2,Int[E^(a+b*x+c*x^2),x]] +
  Dist[1/2,Int[E^(-a-b*x-c*x^2),x]] /;
FreeQ[{a,b,c},x] && NonzeroQ[b^2-4*a*c]
```

$$\int (d+e\,x)^m \operatorname{Sinh}\left[a+b\,x+c\,x^2\right] dx$$

- **Rule: If $b\,e-2\,c\,d=0$, then**

$$\int (d+e\,x) \operatorname{Sinh}\left[a+b\,x+c\,x^2\right] dx \longrightarrow \frac{e \operatorname{Cosh}\left[a+b\,x+c\,x^2\right]}{2\,c}$$

- **Program code:**

```
Int[(d_.+e_.*x_)*Sinh[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  e*Cosh[a+b*x+c*x^2]/(2*c) /;
FreeQ[{a,b,c,d,e},x] && ZeroQ[b*e-2*c*d]
```

```
Int[(d_.+e_.*x_)*Cosh[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  e*Sinh[a+b*x+c*x^2]/(2*c) /;
FreeQ[{a,b,c,d,e},x] && ZeroQ[b*e-2*c*d]
```

- **Rule: If $b\,e-2\,c\,d\neq 0$, then**

$$\int (d+e\,x) \operatorname{Sinh}\left[a+b\,x+c\,x^2\right] dx \longrightarrow \frac{e \operatorname{Cosh}\left[a+b\,x+c\,x^2\right]}{2\,c} - \frac{b\,e-2\,c\,d}{2\,c} \int \operatorname{Sinh}\left[a+b\,x+c\,x^2\right] dx$$

- **Program code:**

```
Int[(d_.+e_.*x_)*Sinh[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  e*Cosh[a+b*x+c*x^2]/(2*c) -
  Dist[(b*e-2*c*d)/(2*c),Int[Sinh[a+b*x+c*x^2],x]] /;
FreeQ[{a,b,c,d,e},x] && NonzeroQ[b*e-2*c*d]
```

```
Int[(d_.+e_.*x_)*Cosh[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  e*Sinh[a+b*x+c*x^2]/(2*c) -
  Dist[(b*e-2*c*d)/(2*c),Int[Cosh[a+b*x+c*x^2],x]] /;
FreeQ[{a,b,c,d,e},x] && NonzeroQ[b*e-2*c*d]
```

- **Rule: If** $m > 1 \land b e - 2 c d = 0$**, then**

$$\int (d+e x)^m \operatorname{Sinh}\left[a+b x+c x^2\right] dx \longrightarrow \frac{e (d+e x)^{m-1} \operatorname{Cosh}\left[a+b x+c x^2\right]}{2 c} + \frac{e^2 (m-1)}{2 c} \int (d+e x)^{m-2} \operatorname{Cosh}\left[a+b x+c x^2\right] dx$$

- **Program code:**

```
Int[(d_.+e_.*x_)^m_*Sinh[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  e*(d+e*x)^(m-1)*Cosh[a+b*x+c*x^2]/(2*c) -
  Dist[e^2*(m-1)/(2*c),Int[(d+e*x)^(m-2)*Cosh[a+b*x+c*x^2],x]] /;
FreeQ[{a,b,c,d,e},x] && RationalQ[m] && m>1 && ZeroQ[b*e-2*c*d]
```

```
Int[(d_.+e_.*x_)^m_*Cosh[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  e*(d+e*x)^(m-1)*Sinh[a+b*x+c*x^2]/(2*c) -
  Dist[e^2*(m-1)/(2*c),Int[(d+e*x)^(m-2)*Sinh[a+b*x+c*x^2],x]] /;
FreeQ[{a,b,c,d,e},x] && RationalQ[m] && m>1 && ZeroQ[b*e-2*c*d]
```

- **Rule: If** $m > 1 \land b e - 2 c d \neq 0$**, then**

$$\int (d+e x)^m \operatorname{Sinh}\left[a+b x+c x^2\right] dx \longrightarrow \frac{e (d+e x)^{m-1} \operatorname{Cosh}\left[a+b x+c x^2\right]}{2 c} - \frac{b e - 2 c d}{2 c} \int (d+e x)^{m-1} \operatorname{Sinh}\left[a+b x+c x^2\right] dx - \frac{e^2 (m-1)}{2 c} \int (d+e x)^{m-2} \operatorname{Cosh}\left[a+b x+c x^2\right] dx$$

- **Program code:**

```
Int[(d_.+e_.*x_)^m_*Sinh[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  e*(d+e*x)^(m-1)*Cosh[a+b*x+c*x^2]/(2*c) -
  Dist[(b*e-2*c*d)/(2*c),Int[(d+e*x)^(m-1)*Sinh[a+b*x+c*x^2],x]] -
  Dist[e^2*(m-1)/(2*c),Int[(d+e*x)^(m-2)*Cosh[a+b*x+c*x^2],x]] /;
FreeQ[{a,b,c,d,e},x] && RationalQ[m] && m>1 && NonzeroQ[b*e-2*c*d]
```

```
Int[(d_.+e_.*x_)^m_*Cosh[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  e*(d+e*x)^(m-1)*Sinh[a+b*x+c*x^2]/(2*c) -
  Dist[(b*e-2*c*d)/(2*c),Int[(d+e*x)^(m-1)*Cosh[a+b*x+c*x^2],x]] -
  Dist[e^2*(m-1)/(2*c),Int[(d+e*x)^(m-2)*Sinh[a+b*x+c*x^2],x]] /;
FreeQ[{a,b,c,d,e},x] && RationalQ[m] && m>1 && NonzeroQ[b*e-2*c*d]
```

- **Rule: If $m < -1 \land b e - 2 c d = 0$, then**

$$\int (d + e x)^m \operatorname{Sinh}\left[a + b x + c x^2\right] dx \longrightarrow \frac{(d + e x)^{m+1} \operatorname{Sinh}\left[a + b x + c x^2\right]}{e (m + 1)} - \frac{2 c}{e^2 (m + 1)} \int (d + e x)^{m+2} \operatorname{Cosh}\left[a + b x + c x^2\right] dx$$

- **Program code:**

```
Int[(d_.+e_.*x_)^m_*Sinh[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  (d+e*x)^(m+1)*Sinh[a+b*x+c*x^2]/(e*(m+1)) -
  Dist[2*c/(e^2*(m+1)),Int[(d+e*x)^(m+2)*Cosh[a+b*x+c*x^2],x]] /;
FreeQ[{a,b,c,d,e},x] && RationalQ[m] && m<-1 && ZeroQ[b*e-2*c*d]
```

```
Int[(d_.+e_.*x_)^m_*Cosh[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  (d+e*x)^(m+1)*Cosh[a+b*x+c*x^2]/(e*(m+1)) -
  Dist[2*c/(e^2*(m+1)),Int[(d+e*x)^(m+2)*Sinh[a+b*x+c*x^2],x]] /;
FreeQ[{a,b,c,d,e},x] && RationalQ[m] && m<-1 && ZeroQ[b*e-2*c*d]
```

- **Rule: If $m < -1 \land b e - 2 c d \neq 0$, then**

$$\int (d + e x)^m \operatorname{Sinh}\left[a + b x + c x^2\right] dx \longrightarrow \frac{(d + e x)^{m+1} \operatorname{Sinh}\left[a + b x + c x^2\right]}{e (m + 1)} - \frac{b e - 2 c d}{e^2 (m + 1)} \int (d + e x)^{m+1} \operatorname{Cosh}\left[a + b x + c x^2\right] dx - \frac{2 c}{e^2 (m + 1)} \int (d + e x)^{m+2} \operatorname{Cosh}\left[a + b x + c x^2\right] dx$$

- **Program code:**

```
Int[(d_.+e_.*x_)^m_*Sinh[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  (d+e*x)^(m+1)*Sinh[a+b*x+c*x^2]/(e*(m+1)) -
  Dist[(b*e-2*c*d)/(e^2*(m+1)),Int[(d+e*x)^(m+1)*Cosh[a+b*x+c*x^2],x]] -
  Dist[2*c/(e^2*(m+1)),Int[(d+e*x)^(m+2)*Cosh[a+b*x+c*x^2],x]] /;
FreeQ[{a,b,c,d,e},x] && RationalQ[m] && m<-1 && NonzeroQ[b*e-2*c*d]
```

```
Int[(d_.+e_.*x_)^m_*Cosh[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  (d+e*x)^(m+1)*Cosh[a+b*x+c*x^2]/(e*(m+1)) -
  Dist[(b*e-2*c*d)/(e^2*(m+1)),Int[(d+e*x)^(m+1)*Sinh[a+b*x+c*x^2],x]] -
  Dist[2*c/(e^2*(m+1)),Int[(d+e*x)^(m+2)*Sinh[a+b*x+c*x^2],x]] /;
FreeQ[{a,b,c,d,e},x] && RationalQ[m] && m<-1 && NonzeroQ[b*e-2*c*d]
```

$$\int \mathrm{Sinh}[a + b\,\mathrm{Log}[c\,x^n]]^p\,dx$$

- **Derivation: Algebraic simplification**
- **Basis:** $\mathrm{Sinh}[b\,\mathrm{Log}[c\,x^n]] = \frac{1}{2}(c\,x^n)^b - \frac{1}{2(c\,x^n)^b}$
- **Rule:**

$$\int \mathrm{Sinh}[b\,\mathrm{Log}[c\,x^n]]^p\,dx \longrightarrow \int\left(\frac{(c\,x^n)^b}{2} - \frac{1}{2(c\,x^n)^b}\right)^p dx$$

- **Program code:**

```
Int[Sinh[b_.*Log[c_.*x_^n_.]]^p_.,x_Symbol] :=
  Int[((c*x^n)^b/2 - 1/(2*(c*x^n)^b))^p,x] /;
FreeQ[c,x] && RationalQ[{b,n,p}]
```

- **Basis:** $\mathrm{Cosh}[b\,\mathrm{Log}[c\,x^n]] = \frac{1}{2}(c\,x^n)^b + \frac{1}{2(c\,x^n)^b}$

```
Int[Cosh[b_.*Log[c_.*x_^n_.]]^p_.,x_Symbol] :=
  Int[((c*x^n)^b/2 + 1/(2*(c*x^n)^b))^p,x] /;
FreeQ[c,x] && RationalQ[{b,n,p}]
```

- **Rule: If** $1 - b^2 n^2 \neq 0$**, then**

$$\int \mathrm{Sinh}[a + b\,\mathrm{Log}[c\,x^n]]\,dx \longrightarrow \frac{x\,\mathrm{Sinh}[a + b\,\mathrm{Log}[c\,x^n]]}{1 - b^2 n^2} - \frac{b\,n\,x\,\mathrm{Cosh}[a + b\,\mathrm{Log}[c\,x^n]]}{1 - b^2 n^2}$$

- **Program code:**

```
Int[Sinh[a_.+b_.*Log[c_.*x_^n_.]],x_Symbol] :=
  x*Sinh[a+b*Log[c*x^n]]/(1-b^2*n^2) -
  b*n*x*Cosh[a+b*Log[c*x^n]]/(1-b^2*n^2) /;
FreeQ[{a,b,c,n},x] && NonzeroQ[1-b^2*n^2]
```

```
Int[Cosh[a_.+b_.*Log[c_.*x_^n_.]],x_Symbol] :=
  x*Cosh[a+b*Log[c*x^n]]/(1-b^2*n^2) -
  b*n*x*Sinh[a+b*Log[c*x^n]]/(1-b^2*n^2) /;
FreeQ[{a,b,c,n},x] && NonzeroQ[1-b^2*n^2]
```

- **Derivation: Piecewise constant extraction**
- **Rule: If $b\,n-2=0$, then**

$$\int\sqrt{\operatorname{Sinh}[a+b\operatorname{Log}[c\,x^n]]}\,dx\;\longrightarrow\;\frac{x\sqrt{\operatorname{Sinh}[a+b\operatorname{Log}[c\,x^n]]}}{\sqrt{-1+e^{2a}\,(c\,x^n)^{4/n}}}\int\frac{\sqrt{-1+e^{2a}\,(c\,x^n)^{4/n}}}{x}\,dx$$

- **Program code:**

```
Int[Sqrt[Sinh[a_.+b_.*Log[c_.*x_^n_.]]],x_Symbol] :=
  x*Sqrt[Sinh[a+b*Log[c*x^n]]]/Sqrt[-1+E^(2*a)*(c*x^n)^(4/n)]*
    Int[Sqrt[-1+E^(2*a)*(c*x^n)^(4/n)]/x,x] /;
FreeQ[{a,b,c,n},x] && ZeroQ[b*n-2]
```

```
Int[Sqrt[Cosh[a_.+b_.*Log[c_.*x_^n_.]]],x_Symbol] :=
  x*Sqrt[Cosh[a+b*Log[c*x^n]]]/Sqrt[1+E^(2*a)*(c*x^n)^(4/n)]*
    Int[Sqrt[1+E^(2*a)*(c*x^n)^(4/n)]/x,x] /;
FreeQ[{a,b,c,n},x] && ZeroQ[b*n-2]
```

- **Rule: If $p>1\;\wedge\;1-b^2\,n^2\,p^2\neq 0$, then**

$$\int\operatorname{Sinh}[a+b\operatorname{Log}[c\,x^n]]^p\,dx\;\longrightarrow\;\frac{x\operatorname{Sinh}[a+b\operatorname{Log}[c\,x^n]]^p}{1-b^2\,n^2\,p^2}-$$
$$\frac{b\,n\,p\,x\operatorname{Cosh}[a+b\operatorname{Log}[c\,x^n]]\operatorname{Sinh}[a+b\operatorname{Log}[c\,x^n]]^{p-1}}{1-b^2\,n^2\,p^2}+\frac{b^2\,n^2\,p\,(p-1)}{1-b^2\,n^2\,p^2}\int\operatorname{Sinh}[a+b\operatorname{Log}[c\,x^n]]^{p-2}\,dx$$

- **Program code:**

```
Int[Sinh[a_.+b_.*Log[c_.*x_^n_.]]^p_,x_Symbol] :=
  x*Sinh[a+b*Log[c*x^n]]^p/(1-b^2*n^2*p^2) -
  b*n*p*x*Cosh[a+b*Log[c*x^n]]*Sinh[a+b*Log[c*x^n]]^(p-1)/(1-b^2*n^2*p^2) +
  Dist[b^2*n^2*p*(p-1)/(1-b^2*n^2*p^2),Int[Sinh[a+b*Log[c*x^n]]^(p-2),x]] /;
FreeQ[{a,b,c,n},x] && RationalQ[p] && p>1 && NonzeroQ[1-b^2*n^2*p^2]
```

```
Int[Cosh[a_.+b_.*Log[c_.*x_^n_.]]^p_,x_Symbol] :=
  x*Cosh[a+b*Log[c*x^n]]^p/(1-b^2*n^2*p^2) -
  b*n*p*x*Sinh[a+b*Log[c*x^n]]*Cosh[a+b*Log[c*x^n]]^(p-1)/(1-b^2*n^2*p^2) -
  Dist[b^2*n^2*p*(p-1)/(1-b^2*n^2*p^2),Int[Cosh[a+b*Log[c*x^n]]^(p-2),x]] /;
FreeQ[{a,b,c,n},x] && RationalQ[p] && p>1 && NonzeroQ[1-b^2*n^2*p^2]
```

- **Rule: If** $p \neq -1 \land p \neq -2 \land 1 - b^2 n^2 (p+2)^2 = 0$, **then**

$$\int \text{Sinh}[a + b\,\text{Log}[c\,x^n]]^p\,dx \longrightarrow \frac{x\,\text{Coth}[a + b\,\text{Log}[c\,x^n]]\,\text{Sinh}[a + b\,\text{Log}[c\,x^n]]^{p+2}}{b\,n\,(p+1)} - \frac{x\,\text{Sinh}[a + b\,\text{Log}[c\,x^n]]^{p+2}}{b^2\,n^2\,(p+1)\,(p+2)}$$

- **Program code:**

```
Int[Sinh[a_.+b_.*Log[c_.*x_^n_.]]^p_,x_Symbol] :=
  x*Coth[a+b*Log[c*x^n]]*Sinh[a+b*Log[c*x^n]]^(p+2)/(b*n*(p+1)) -
  x*Sinh[a+b*Log[c*x^n]]^(p+2)/(b^2*n^2*(p+1)*(p+2)) /;
FreeQ[{a,b,c,n,p},x] && NonzeroQ[p+1] && NonzeroQ[p+2] && ZeroQ[1-b^2*n^2*(p+2)^2]
```

```
Int[Cosh[a_.+b_.*Log[c_.*x_^n_.]]^p_,x_Symbol] :=
  -x*Tanh[a+b*Log[c*x^n]]*Cosh[a+b*Log[c*x^n]]^(p+2)/(b*n*(p+1)) +
  x*Cosh[a+b*Log[c*x^n]]^(p+2)/(b^2*n^2*(p+1)*(p+2)) /;
FreeQ[{a,b,c,n,p},x] && NonzeroQ[p+1] && NonzeroQ[p+2] && ZeroQ[1-b^2*n^2*(p+2)^2]
```

- **Rule: If** $p < -1 \land p \neq -2 \land 1 - b^2 n^2 (p+2)^2 \neq 0$, **then**

$$\int \text{Sinh}[a + b\,\text{Log}[c\,x^n]]^p\,dx \longrightarrow \frac{x\,\text{Coth}[a + b\,\text{Log}[c\,x^n]]\,\text{Sinh}[a + b\,\text{Log}[c\,x^n]]^{p+2}}{b\,n\,(p+1)} - \frac{x\,\text{Sinh}[a + b\,\text{Log}[c\,x^n]]^{p+2}}{b^2\,n^2\,(p+1)\,(p+2)} + \frac{1 - b^2\,n^2\,(p+2)^2}{b^2\,n^2\,(p+1)\,(p+2)} \int \text{Sinh}[a + b\,\text{Log}[c\,x^n]]^{p+2}\,dx$$

- **Program code:**

```
Int[Sinh[a_.+b_.*Log[c_.*x_^n_.]]^p_,x_Symbol] :=
  x*Coth[a+b*Log[c*x^n]]*Sinh[a+b*Log[c*x^n]]^(p+2)/(b*n*(p+1)) -
  x*Sinh[a+b*Log[c*x^n]]^(p+2)/(b^2*n^2*(p+1)*(p+2)) +
  Dist[(1-b^2*n^2*(p+2)^2)/(b^2*n^2*(p+1)*(p+2)),Int[Sinh[a+b*Log[c*x^n]]^(p+2),x]] /;
FreeQ[{a,b,c,n},x] && RationalQ[p] && p<-1 && p!=-2 && NonzeroQ[1-b^2*n^2*(p+2)^2]
```

```
Int[Cosh[a_.+b_.*Log[c_.*x_^n_.]]^p_,x_Symbol] :=
  -x*Tanh[a+b*Log[c*x^n]]*Cosh[a+b*Log[c*x^n]]^(p+2)/(b*n*(p+1)) +
  x*Cosh[a+b*Log[c*x^n]]^(p+2)/(b^2*n^2*(p+1)*(p+2)) -
  Dist[(1-b^2*n^2*(p+2)^2)/(b^2*n^2*(p+1)*(p+2)),Int[Cosh[a+b*Log[c*x^n]]^(p+2),x]] /;
FreeQ[{a,b,c,n},x] && RationalQ[p] && p<-1 && p≠-2 && NonzeroQ[1-b^2*n^2*(p+2)^2]
```

$$\int x^m \operatorname{Sinh}[a + b \operatorname{Log}[c\, x^n]]^p \, dx$$

- **Rule: If** $(m+1)^2 - b^2 n^2 \neq 0 \land m+1 \neq 0$**, then**

$$\int x^m \operatorname{Sinh}[a + b \operatorname{Log}[c\, x^n]] \, dx \longrightarrow \frac{(m+1)\, x^{m+1} \operatorname{Sinh}[a + b \operatorname{Log}[c\, x^n]]}{(m+1)^2 - b^2 n^2} - \frac{b\, n\, x^{m+1} \operatorname{Cosh}[a + b \operatorname{Log}[c\, x^n]]}{(m+1)^2 - b^2 n^2}$$

- **Program code:**

```
Int[x_^m_.*Sinh[a_.+b_.*Log[c_.*x_^n_.]],x_Symbol] :=
  (m+1)*x^(m+1)*Sinh[a+b*Log[c*x^n]]/((m+1)^2-b^2*n^2) -
  b*n*x^(m+1)*Cosh[a+b*Log[c*x^n]]/((m+1)^2-b^2*n^2) /;
FreeQ[{a,b,c,m,n},x] && NonzeroQ[(m+1)^2-b^2*n^2] && NonzeroQ[m+1]
```

```
Int[x_^m_.*Cosh[a_.+b_.*Log[c_.*x_^n_.]],x_Symbol] :=
  (m+1)*x^(m+1)*Cosh[a+b*Log[c*x^n]]/((m+1)^2-b^2*n^2) -
  b*n*x^(m+1)*Sinh[a+b*Log[c*x^n]]/((m+1)^2-b^2*n^2) /;
FreeQ[{a,b,c,m,n},x] && NonzeroQ[(m+1)^2-b^2*n^2] && NonzeroQ[m+1]
```

- **Rule: If** $(m+1)^2 - b^2 n^2 p^2 \neq 0 \land p > 1 \land m+1 \neq 0$**, then**

$$\int x^m \operatorname{Sinh}[a + b \operatorname{Log}[c\, x^n]]^p \, dx \longrightarrow \frac{(m+1)\, x^{m+1} \operatorname{Sinh}[a + b \operatorname{Log}[c\, x^n]]^p}{(m+1)^2 - b^2 n^2 p^2} -$$

$$\frac{b\, n\, p\, x^{m+1} \operatorname{Cosh}[a + b \operatorname{Log}[c\, x^n]] \operatorname{Sinh}[a + b \operatorname{Log}[c\, x^n]]^{p-1}}{(m+1)^2 - b^2 n^2 p^2} + \frac{b^2 n^2 p\, (p-1)}{(m+1)^2 - b^2 n^2 p^2} \int x^m \operatorname{Sinh}[a + b \operatorname{Log}[c\, x^n]]^{p-2} \, dx$$

- **Program code:**

```
Int[x_^m_.*Sinh[a_.+b_.*Log[c_.*x_^n_.]]^p_,x_Symbol] :=
  (m+1)*x^(m+1)*Sinh[a+b*Log[c*x^n]]^p/((m+1)^2-b^2*n^2*p^2) -
  b*n*p*x^(m+1)*Cosh[a+b*Log[c*x^n]]*Sinh[a+b*Log[c*x^n]]^(p-1)/((m+1)^2-b^2*n^2*p^2) +
  Dist[b^2*n^2*p*(p-1)/((m+1)^2-b^2*n^2*p^2),Int[x^m*Sinh[a+b*Log[c*x^n]]^(p-2),x]] /;
FreeQ[{a,b,c,m,n},x] && NonzeroQ[(m+1)^2-b^2*n^2*p^2] && RationalQ[p] && p>1 && NonzeroQ[m+1]
```

```
Int[x_^m_.*Cosh[a_.+b_.*Log[c_.*x_^n_.]]^p_,x_Symbol] :=
  (m+1)*x^(m+1)*Cosh[a+b*Log[c*x^n]]^p/((m+1)^2-b^2*n^2*p^2) -
  b*n*p*x^(m+1)*Cosh[a+b*Log[c*x^n]]^(p-1)*Sinh[a+b*Log[c*x^n]]/((m+1)^2-b^2*n^2*p^2) -
  Dist[b^2*n^2*p*(p-1)/((m+1)^2-b^2*n^2*p^2),Int[x^m*Cosh[a+b*Log[c*x^n]]^(p-2),x]] /;
FreeQ[{a,b,c,m,n},x] && NonzeroQ[(m+1)^2-b^2*n^2*p^2] && RationalQ[p] && p>1 && NonzeroQ[m+1]
```

- **Rule: If** $(m+1)^2 - b^2 n^2 (p+2)^2 = 0 \wedge p \neq -1 \wedge p \neq -2$, **then**

$$\int x^m \operatorname{Sinh}[a+b\operatorname{Log}[c\,x^n]]^p\,dx \longrightarrow \frac{x^{m+1}\operatorname{Coth}[a+b\operatorname{Log}[c\,x^n]]\operatorname{Sinh}[a+b\operatorname{Log}[c\,x^n]]^{p+2}}{b\,n\,(p+1)} - \frac{(m+1)\,x^{m+1}\operatorname{Sinh}[a+b\operatorname{Log}[c\,x^n]]^{p+2}}{b^2 n^2 (p+1)(p+2)}$$

- **Program code:**

```
Int[x_^m_.*Sinh[a_.+b_.*Log[c_.*x_^n_.]]^p_,x_Symbol] :=
  x^(m+1)*Coth[a+b*Log[c*x^n]]*Sinh[a+b*Log[c*x^n]]^(p+2)/(b*n*(p+1)) -
  (m+1)*x^(m+1)*Sinh[a+b*Log[c*x^n]]^(p+2)/(b^2*n^2*(p+1)*(p+2)) /;
FreeQ[{a,b,c,m,n,p},x] && ZeroQ[(m+1)^2-b^2*n^2*(p+2)^2] && NonzeroQ[p+1] && NonzeroQ[p+2]
```

```
Int[x_^m_.*Cosh[a_.+b_.*Log[c_.*x_^n_.]]^p_,x_Symbol] :=
  -x^(m+1)*Tanh[a+b*Log[c*x^n]]*Cosh[a+b*Log[c*x^n]]^(p+2)/(b*n*(p+1)) +
  (m+1)*x^(m+1)*Cosh[a+b*Log[c*x^n]]^(p+2)/(b^2*n^2*(p+1)*(p+2)) /;
FreeQ[{a,b,c,m,n,p},x] && ZeroQ[(m+1)^2-b^2*n^2*(p+2)^2] && NonzeroQ[p+1] && NonzeroQ[p+2]
```

- **Rule: If** $(m+1)^2 - b^2 n^2 (p+2)^2 \neq 0 \wedge p < -1 \wedge p \neq -2 \wedge m+1 \neq 0$, **then**

$$\int x^m \operatorname{Sinh}[a+b\operatorname{Log}[c\,x^n]]^p\,dx \longrightarrow \frac{x^{m+1}\operatorname{Coth}[a+b\operatorname{Log}[c\,x^n]]\operatorname{Sinh}[a+b\operatorname{Log}[c\,x^n]]^{p+2}}{b\,n\,(p+1)} - \frac{(m+1)\,x^{m+1}\operatorname{Sinh}[a+b\operatorname{Log}[c\,x^n]]^{p+2}}{b^2 n^2 (p+1)(p+2)} + \frac{(m+1)^2 - b^2 n^2 (p+2)^2}{b^2 n^2 (p+1)(p+2)} \int x^m \operatorname{Sinh}[a+b\operatorname{Log}[c\,x^n]]^{p+2}\,dx$$

- **Program code:**

```
Int[x_^m_.*Sinh[a_.+b_.*Log[c_.*x_^n_.]]^p_,x_Symbol] :=
  x^(m+1)*Coth[a+b*Log[c*x^n]]*Sinh[a+b*Log[c*x^n]]^(p+2)/(b*n*(p+1)) -
  (m+1)*x^(m+1)*Sinh[a+b*Log[c*x^n]]^(p+2)/(b^2*n^2*(p+1)*(p+2)) +
  Dist[((m+1)^2-b^2*n^2*(p+2)^2)/(b^2*n^2*(p+1)*(p+2)),Int[x^m*Sinh[a+b*Log[c*x^n]]^(p+2),x]] /;
FreeQ[{a,b,c,m,n},x] && NonzeroQ[(m+1)^2-b^2*n^2*(p+2)^2] && RationalQ[p] && p<-1 && p!=-2 && Nonzer
```

```
Int[x_^m_.*Cosh[a_.+b_.*Log[c_.*x_^n_.]]^p_,x_Symbol] :=
  -x^(m+1)*Tanh[a+b*Log[c*x^n]]*Cosh[a+b*Log[c*x^n]]^(p+2)/(b*n*(p+1)) +
  (m+1)*x^(m+1)*Cosh[a+b*Log[c*x^n]]^(p+2)/(b^2*n^2*(p+1)*(p+2)) -
  Dist[((m+1)^2-b^2*n^2*(p+2)^2)/(b^2*n^2*(p+1)*(p+2)),Int[x^m*Cosh[a+b*Log[c*x^n]]^(p+2),x]] /;
FreeQ[{a,b,c,m,n},x] && NonzeroQ[(m+1)^2-b^2*n^2*(p+2)^2] && RationalQ[p] && p<-1 && p≠-2 && Nonzero
```

$$\int \mathrm{Sinh}[a\, x^n\, \mathrm{Log}[b\, x]^p]\, \mathrm{Log}[b\, x]^p\, dx$$

- **Rule: If $p > 0$, then**

$$\int \mathrm{Sinh}[a\, x\, \mathrm{Log}[b\, x]^p]\, \mathrm{Log}[b\, x]^p\, dx \longrightarrow \frac{\mathrm{Cosh}[a\, x\, \mathrm{Log}[b\, x]^p]}{a} - p \int \mathrm{Sinh}[a\, x\, \mathrm{Log}[b\, x]^p]\, \mathrm{Log}[b\, x]^{p-1}\, dx$$

- **Program code:**

```
Int[Sinh[a_.*x_*Log[b_.*x_]^p_.]*Log[b_.*x_]^p_.,x_Symbol] :=
  Cosh[a*x*Log[b*x]^p]/a -
  Dist[p,Int[Sinh[a*x*Log[b*x]^p]*Log[b*x]^(p-1),x]] /;
FreeQ[{a,b},x] && RationalQ[p] && p>0
```

```
Int[Cosh[a_.*x_*Log[b_.*x_]^p_.]*Log[b_.*x_]^p_.,x_Symbol] :=
  Sinh[a*x*Log[b*x]^p]/a -
  Dist[p,Int[Cosh[a*x*Log[b*x]^p]*Log[b*x]^(p-1),x]] /;
FreeQ[{a,b},x] && RationalQ[p] && p>0
```

- **Rule: If $p > 0$, then**

$$\int \mathrm{Sinh}[a\, x^n\, \mathrm{Log}[b\, x]^p]\, \mathrm{Log}[b\, x]^p\, dx \longrightarrow \frac{\mathrm{Cosh}[a\, x^n\, \mathrm{Log}[b\, x]^p]}{a\, n\, x^{n-1}} -$$
$$\frac{p}{n} \int \mathrm{Sinh}[a\, x^n\, \mathrm{Log}[b\, x]^p]\, \mathrm{Log}[b\, x]^{p-1}\, dx + \frac{n-1}{a\, n} \int \frac{\mathrm{Cosh}[a\, x^n\, \mathrm{Log}[b\, x]^p]}{x^n}\, dx$$

- **Program code:**

```
Int[Sinh[a_.*x_^n_*Log[b_.*x_]^p_.]*Log[b_.*x_]^p_.,x_Symbol] :=
  Cosh[a*x^n*Log[b*x]^p]/(a*n*x^(n-1)) -
  Dist[p/n,Int[Sinh[a*x^n*Log[b*x]^p]*Log[b*x]^(p-1),x]] +
  Dist[(n-1)/(a*n),Int[Cosh[a*x^n*Log[b*x]^p]/x^n,x]] /;
FreeQ[{a,b},x] && RationalQ[{n,p}] && p>0
```

```
Int[Cosh[a_.*x_^n_*Log[b_.*x_]^p_.]*Log[b_.*x_]^p_.,x_Symbol] :=
  Sinh[a*x^n*Log[b*x]^p]/(a*n*x^(n-1)) -
  Dist[p/n,Int[Cosh[a*x^n*Log[b*x]^p]*Log[b*x]^(p-1),x]] +
  Dist[(n-1)/(a*n),Int[Sinh[a*x^n*Log[b*x]^p]/x^n,x]] /;
FreeQ[{a,b},x] && RationalQ[{n,p}] && p>0
```

$$\int x^m \operatorname{Sinh}\left[a x^n \operatorname{Log}[b x]^p\right] \operatorname{Log}[b x]^p \, dx$$

- **Rule: If $p > 0 \wedge m - n + 1 = 0$, then**

$$\int x^m \operatorname{Sinh}\left[a x^n \operatorname{Log}[b x]^p\right] \operatorname{Log}[b x]^p \, dx \longrightarrow \frac{\operatorname{Cosh}\left[a x^n \operatorname{Log}[b x]^p\right]}{a n} - \frac{p}{n} \int x^m \operatorname{Sinh}\left[a x^n \operatorname{Log}[b x]^p\right] \operatorname{Log}[b x]^{p-1} \, dx$$

- **Program code:**

```
Int[x_^m_.*Sinh[a_.*x_^n_.*Log[b_.*x_]^p_.]*Log[b_.*x_]^p_.,x_Symbol] :=
  Cosh[a*x^n*Log[b*x]^p]/(a*n) -
  Dist[p/n,Int[x^m*Sinh[a*x^n*Log[b*x]^p]*Log[b*x]^(p-1),x]] /;
FreeQ[{a,b},x] && RationalQ[{m,n,p}] && p>0 && ZeroQ[m-n+1]
```

```
Int[x_^m_.*Cosh[a_.*x_^n_.*Log[b_.*x_]^p_.]*Log[b_.*x_]^p_.,x_Symbol] :=
  Sinh[a*x^n*Log[b*x]^p]/(a*n) -
  Dist[p/n,Int[x^m*Cosh[a*x^n*Log[b*x]^p]*Log[b*x]^(p-1),x]] /;
FreeQ[{a,b},x] && RationalQ[{m,n,p}] && p>0 && ZeroQ[m-n+1]
```

- **Rule: If $p > 0 \wedge m - n + 1 \neq 0$, then**

$$\int x^m \operatorname{Sinh}\left[a x^n \operatorname{Log}[b x]^p\right] \operatorname{Log}[b x]^p \, dx \longrightarrow \frac{x^{m-n+1} \operatorname{Cosh}\left[a x^n \operatorname{Log}[b x]^p\right]}{a n} - \frac{p}{n} \int x^m \operatorname{Sinh}\left[a x^n \operatorname{Log}[b x]^p\right] \operatorname{Log}[b x]^{p-1} \, dx - \frac{m-n+1}{a n} \int x^{m-n} \operatorname{Cosh}\left[a x^n \operatorname{Log}[b x]^p\right] \, dx$$

- **Program code:**

```
Int[x_^m_*Sinh[a_.*x_^n_.*Log[b_.*x_]^p_.]*Log[b_.*x_]^p_.,x_Symbol] :=
  x^(m-n+1)*Cosh[a*x^n*Log[b*x]^p]/(a*n) -
  Dist[p/n,Int[x^m*Sinh[a*x^n*Log[b*x]^p]*Log[b*x]^(p-1),x]] -
  Dist[(m-n+1)/(a*n),Int[x^(m-n)*Cosh[a*x^n*Log[b*x]^p],x]] /;
FreeQ[{a,b},x] && RationalQ[{m,n,p}] && p>0 && NonzeroQ[m-n+1]
```

```
Int[x_^m_*Cosh[a_.*x_^n_.*Log[b_.*x_]^p_.]*Log[b_.*x_]^p_.,x_Symbol] :=
  x^(m-n+1)*Sinh[a*x^n*Log[b*x]^p]/(a*n) -
  Dist[p/n,Int[x^m*Cosh[a*x^n*Log[b*x]^p]*Log[b*x]^(p-1),x]] -
  Dist[(m-n+1)/(a*n),Int[x^(m-n)*Sinh[a*x^n*Log[b*x]^p],x]] /;
FreeQ[{a,b},x] && RationalQ[{m,n,p}] && p>0 && NonzeroQ[m-n+1]
```

$\int u \,\text{Sinh}[a + b\,x]^n \,dx$

- **Derivation: Algebraic expansion**
- **Basis:** $\text{Sinh}[z]^2 = -\frac{1}{2} + \frac{1}{2}\,\text{Cosh}[2\,z]$
- **Rule: If** $\frac{n-1}{2} \notin \mathbb{Z}$**, then**

$$\int \text{Sinh}\left[\frac{a}{2} + \frac{b\,x}{2}\right]^2 \text{Sinh}[a + b\,x]^n \,dx \longrightarrow -\frac{1}{2}\int \text{Sinh}[a + b\,x]^n \,dx + \frac{1}{2}\int \text{Cosh}[a + b\,x]\,\text{Sinh}[a + b\,x]^n \,dx$$

- **Program code:**

```
Int[Sinh[c_.+d_.*x_]^2*Sinh[a_.+b_.*x_]^n_.,x_Symbol] :=
  -Dist[1/2,Int[Sinh[a+b*x]^n,x]] +
  Dist[1/2,Int[Cosh[a+b*x]*Sinh[a+b*x]^n,x]] /;
FreeQ[{a,b,c,d},x] && ZeroQ[c-a/2] && ZeroQ[d-b/2] && Not[OddQ[n]]
```

- **Derivation: Algebraic expansion**
- **Basis:** $\text{Cosh}[z]^2 = \frac{1}{2} + \frac{1}{2}\,\text{Cosh}[2\,z]$
- **Rule: If** $\frac{n-1}{2} \notin \mathbb{Z}$**, then**

$$\int \text{Cosh}\left[\frac{a}{2} + \frac{b\,x}{2}\right]^2 \text{Sinh}[a + b\,x]^n \,dx \longrightarrow \frac{1}{2}\int \text{Sinh}[a + b\,x]^n \,dx + \frac{1}{2}\int \text{Cosh}[a + b\,x]\,\text{Sinh}[a + b\,x]^n \,dx$$

- **Program code:**

```
Int[Cosh[c_.+d_.*x_]^2*Sinh[a_.+b_.*x_]^n_.,x_Symbol] :=
  Dist[1/2,Int[Sinh[a+b*x]^n,x]] +
  Dist[1/2,Int[Cosh[a+b*x]*Sinh[a+b*x]^n,x]] /;
FreeQ[{a,b,c,d},x] && ZeroQ[c-a/2] && ZeroQ[d-b/2] && Not[OddQ[n]]
```

- **Derivation: Algebraic simplification**
- **Basis:** $\text{Sinh}[2z] = 2\,\text{Sinh}[z]\,\text{Cosh}[z]$
- **Rule: If** $n \in \mathbb{Z}$ **and** u **is a function of trig functions of** $\frac{a}{2} + \frac{b x}{2}$**, then**

$$\int u\,\text{Sinh}[a + b x]^n\,dx \longrightarrow 2^n \int u\,\text{Cosh}\left[\frac{a}{2} + \frac{b x}{2}\right]^n \text{Sinh}\left[\frac{a}{2} + \frac{b x}{2}\right]^n dx$$

- **Program code:**

```
Int[u_*Sinh[a_.+b_.*x_]^n_.,x_Symbol] :=
  Dist[2^n,Int[u*Cosh[a/2+b*x/2]^n*Sinh[a/2+b*x/2]^n,x]] /;
FreeQ[{a,b},x] && IntegerQ[n] && ZeroQ[a/2+b*x/2-FunctionOfHyperbolic[u,x]]
```

$$\int u \,\mathrm{Sinh}[v]^2 \,dx$$

- **Derivation: Algebraic expansion**

- **Basis:** $\mathrm{Sinh}[z]^2 = -\frac{1}{2} + \frac{1}{2}\mathrm{Cosh}[2z]$

- **Rule: If u is a function of hyperbolic functions of 2 v, then**

$$\int u \,\mathrm{Sinh}[v]^2 \,dx \longrightarrow -\frac{1}{2}\int u \,dx + \frac{1}{2}\int u \,\mathrm{Cosh}[2v] \,dx$$

- **Program code:**

```
Int[u_*Sinh[v_]^2,x_Symbol] :=
  Dist[-1/2,Int[u,x]] +
  Dist[1/2,Int[u*Cosh[2*v],x]] /;
FunctionOfHyperbolicQ[u,2*v,x]
```

```
Int[u_*Cosh[v_]^2,x_Symbol] :=
  Dist[1/2,Int[u,x]] +
  Dist[1/2,Int[u*Cosh[2*v],x]] /;
FunctionOfHyperbolicQ[u,2*v,x]
```