$$\int \mathrm{Erf}[a+b\,x]^n\,dx$$

- **Reference: G&R 5.41**
- **Derivation: Integration by parts**
- **Rule:**

$$\int \mathrm{Erf}[a+b\,x]\,dx \longrightarrow \frac{(a+b\,x)\,\mathrm{Erf}[a+b\,x]}{b}+\frac{1}{b\,\sqrt{\pi}\,e^{(a+b\,x)^2}}$$

- **Program code:**

```
Int[Erf[a_.+b_.*x_],x_Symbol] :=
  (a+b*x)*Erf[a+b*x]/b + 1/(b*Sqrt[Pi]*E^(a+b*x)^2) /;
FreeQ[{a,b},x]
```

- **Derivation: Integration by parts**
- **Rule:**

$$\int \mathrm{Erf}[a+b\,x]^2\,dx \longrightarrow \frac{(a+b\,x)\,\mathrm{Erf}[a+b\,x]^2}{b}-\frac{4}{\sqrt{\pi}}\int \frac{(a+b\,x)\,\mathrm{Erf}[a+b\,x]}{e^{(a+b\,x)^2}}\,dx$$

- **Program code:**

```
Int[Erf[a_.+b_.*x_]^2,x_Symbol] :=
  (a+b*x)*Erf[a+b*x]^2/b -
  Dist[4/Sqrt[Pi],Int[(a+b*x)*Erf[a+b*x]/E^(a+b*x)^2,x]] /;
FreeQ[{a,b},x]
```

$$\int x^m \operatorname{Erf}[a+b x]^n \, dx$$

- **Derivation: Integration by parts**
- **Rule: If** $m+1 \neq 0$, **then**

$$\int x^m \operatorname{Erf}[a+b x] \, dx \longrightarrow \frac{x^{m+1} \operatorname{Erf}[a+b x]}{m+1} - \frac{2 b}{\sqrt{\pi}\,(m+1)} \int \frac{x^{m+1}}{e^{(a+b x)^2}} \, dx$$

- **Program code:**

```
Int[x_^m_.*Erf[a_.+b_.*x_],x_Symbol] :=
  x^(m+1)*Erf[a+b*x]/(m+1) -
  Dist[2*b/(Sqrt[Pi]*(m+1)),Int[x^(m+1)/E^(a+b*x)^2,x]] /;
FreeQ[{a,b,m},x] && NonzeroQ[m+1]
```

- **Derivation: Integration by parts**
- **Rule: If** $m \in \mathbb{Z} \bigwedge m+1 \neq 0 \bigwedge \left(m>0 \bigvee \frac{m-1}{2} \in \mathbb{Z}\right)$, **then**

$$\int x^m \operatorname{Erf}[b x]^2 \, dx \longrightarrow \frac{x^{m+1} \operatorname{Erf}[b x]^2}{m+1} - \frac{4 b}{\sqrt{\pi}\,(m+1)} \int \frac{x^{m+1} \operatorname{Erf}[b x]}{e^{b^2 x^2}} \, dx$$

- **Program code:**

```
Int[x_^m_.*Erf[b_.*x_]^2,x_Symbol] :=
  x^(m+1)*Erf[b*x]^2/(m+1) -
  Dist[4*b/(Sqrt[Pi]*(m+1)),Int[x^(m+1)*E^(-b^2*x^2)*Erf[b*x],x]] /;
FreeQ[b,x] && IntegerQ[m] && m+1!=0 && (m>0 || OddQ[m])
```

- **Derivation: Integration by substitution**
- **Basis:** $x^m f[a+b x] = \frac{1}{b}\left(-\frac{a}{b}+\frac{a+b x}{b}\right)^m f[a+b x]\, \partial_x (a+b x)$
- **Rule: If** $m \in \mathbb{Z} \bigwedge m>0$, **then**

$$\int x^m \operatorname{Erf}[a+b x]^2 \, dx \longrightarrow \frac{1}{b} \operatorname{Subst}\left[\int \left(-\frac{a}{b}+\frac{x}{b}\right)^m \operatorname{Erf}[x]^2 \, dx,\ x,\ a+b x\right]$$

- **Program code:**

```
Int[x_^m_.*Erf[a_+b_.*x_]^2,x_Symbol] :=
  Dist[1/b,Subst[Int[(-a/b+x/b)^m*Erf[x]^2,x],x,a+b*x]] /;
FreeQ[{a,b},x] && IntegerQ[m] && m>0
```

$$\int \frac{x^m \,\mathrm{Erf}[b\,x]}{e^{b^2 x^2}}\,dx$$

- **Derivation: Integration by parts special case**
- **Rule:**

$$\int \frac{x\,\mathrm{Erf}[b\,x]}{e^{b^2 x^2}}\,dx \longrightarrow -\frac{\mathrm{Erf}[b\,x]}{2\,b^2\,e^{b^2 x^2}} + \frac{1}{b\sqrt{\pi}}\int \frac{1}{e^{2\,b^2 x^2}}\,dx$$

- **Program code:**

```
Int[x_*E^(c_.*x_^2)*Erf[b_.*x_],x_Symbol] :=
  -E^(-b^2*x^2)*Erf[b*x]/(2*b^2) +
  Dist[1/(b*Sqrt[Pi]),Int[E^(-2*b^2*x^2),x]] /;
FreeQ[{b,c},x] && ZeroQ[c+b^2]
```

- **Derivation: Integration by parts**
- **Rule: If** $m \in \mathbb{Z} \bigwedge m > 1$**, then**

$$\int \frac{x^m \,\mathrm{Erf}[b\,x]}{e^{b^2 x^2}}\,dx \longrightarrow -\frac{x^{m-1}\,\mathrm{Erf}[b\,x]}{2\,b^2\,e^{b^2 x^2}} + \frac{1}{b\sqrt{\pi}}\int \frac{x^{m-1}}{e^{2\,b^2 x^2}}\,dx + \frac{m-1}{2\,b^2}\int \frac{x^{m-2}\,\mathrm{Erf}[b\,x]}{e^{b^2 x^2}}\,dx$$

- **Program code:**

```
Int[x_^m_*E^(c_.*x_^2)*Erf[b_.*x_],x_Symbol] :=
  -x^(m-1)*E^(-b^2*x^2)*Erf[b*x]/(2*b^2) +
  Dist[1/(b*Sqrt[Pi]),Int[x^(m-1)*E^(-2*b^2*x^2),x]] +
  Dist[(m-1)/(2*b^2),Int[x^(m-2)*E^(-b^2*x^2)*Erf[b*x],x]] /;
FreeQ[{b,c},x] && ZeroQ[c+b^2] && IntegerQ[m] && m>1
```

- **Derivation: Inverted integration by parts**
- **Rule: If** $\frac{m}{2} \in \mathbb{Z} \bigwedge m < -1$**, then**

$$\int \frac{x^m \,\mathrm{Erf}[b\,x]}{e^{b^2 x^2}}\,dx \longrightarrow \frac{x^{m+1}\,\mathrm{Erf}[b\,x]}{e^{b^2 x^2}\,(m+1)} - \frac{2\,b}{\sqrt{\pi}\,(m+1)}\int \frac{x^{m+1}}{e^{2\,b^2 x^2}}\,dx + \frac{2\,b^2}{m+1}\int \frac{x^{m+2}\,\mathrm{Erf}[b\,x]}{e^{b^2 x^2}}\,dx$$

- **Program code:**

```
Int[x_^m_*E^(c_.*x_^2)*Erf[b_.*x_],x_Symbol] :=
  x^(m+1)*E^(-b^2*x^2)*Erf[b*x]/(m+1) -
  Dist[2*b/(Sqrt[Pi]*(m+1)),Int[x^(m+1)*E^(-2*b^2*x^2),x]] +
  Dist[2*b^2/(m+1),Int[x^(m+2)*E^(-b^2*x^2)*Erf[b*x],x]] /;
FreeQ[{b,c},x] && ZeroQ[c+b^2] && EvenQ[m] && m<-1
```

$$\int \mathrm{Erfc}[a+b\,x]^n\,dx$$

- **Derivation: Integration by parts**
- **Rule:**

$$\int \mathrm{Erfc}[a+b\,x]\,dx \longrightarrow \frac{(a+b\,x)\,\mathrm{Erfc}[a+b\,x]}{b}-\frac{1}{b\,\sqrt{\pi}\,e^{(a+b\,x)^2}}$$

- **Program code:**

```
Int[Erfc[a_.+b_.*x_],x_Symbol] :=
  (a+b*x)*Erfc[a+b*x]/b - 1/(b*Sqrt[Pi]*E^(a+b*x)^2) /;
FreeQ[{a,b},x]
```

- **Derivation: Integration by parts**
- **Rule:**

$$\int \mathrm{Erfc}[a+b\,x]^2\,dx \longrightarrow \frac{(a+b\,x)\,\mathrm{Erfc}[a+b\,x]^2}{b}+\frac{4}{\sqrt{\pi}}\int \frac{(a+b\,x)\,\mathrm{Erfc}[a+b\,x]}{e^{(a+b\,x)^2}}\,dx$$

- **Program code:**

```
Int[Erfc[a_.+b_.*x_]^2,x_Symbol] :=
  (a+b*x)*Erfc[a+b*x]^2/b +
  Dist[4/Sqrt[Pi],Int[(a+b*x)*Erfc[a+b*x]/E^(a+b*x)^2,x]] /;
FreeQ[{a,b},x]
```

$$\int x^m \operatorname{Erfc}[a+b\,x]^n \, dx$$

- **Derivation: Integration by parts**
- **Rule: If $m+1 \neq 0$, then**

$$\int x^m \operatorname{Erfc}[a+b\,x] \, dx \longrightarrow \frac{x^{m+1} \operatorname{Erfc}[a+b\,x]}{m+1} + \frac{2\,b}{\sqrt{\pi}\,(m+1)} \int \frac{x^{m+1}}{e^{(a+b\,x)^2}} \, dx$$

- **Program code:**

```
Int[x_^m_.*Erfc[a_.+b_.*x_],x_Symbol] :=
  x^(m+1)*Erfc[a+b*x]/(m+1) +
  Dist[2*b/(Sqrt[Pi]*(m+1)),Int[x^(m+1)/E^(a+b*x)^2,x]] /;
FreeQ[{a,b,m},x] && NonzeroQ[m+1]
```

- **Derivation: Integration by parts**
- **Rule: If $m \in \mathbb{Z} \bigwedge m+1 \neq 0 \bigwedge \left(m > 0 \bigvee \frac{m-1}{2} \in \mathbb{Z}\right)$, then**

$$\int x^m \operatorname{Erfc}[b\,x]^2 \, dx \longrightarrow \frac{x^{m+1} \operatorname{Erfc}[b\,x]^2}{m+1} + \frac{4\,b}{\sqrt{\pi}\,(m+1)} \int \frac{x^{m+1} \operatorname{Erfc}[b\,x]}{e^{b^2\,x^2}} \, dx$$

- **Program code:**

```
Int[x_^m_.*Erfc[b_.*x_]^2,x_Symbol] :=
  x^(m+1)*Erfc[b*x]^2/(m+1) +
  Dist[4*b/(Sqrt[Pi]*(m+1)),Int[x^(m+1)*E^(-b^2*x^2)*Erfc[b*x],x]] /;
FreeQ[b,x] && IntegerQ[m] && m+1!=0 && (m>0 || OddQ[m])
```

- **Derivation: Integration by substitution**
- **Basis:** $x^m f[a+b\,x] = \frac{1}{b}\left(-\frac{a}{b}+\frac{a+b\,x}{b}\right)^m f[a+b\,x]\, \partial_x (a+b\,x)$
- **Rule: If $m \in \mathbb{Z} \bigwedge m > 0$, then**

$$\int x^m \operatorname{Erfc}[a+b\,x]^2 \, dx \longrightarrow \frac{1}{b} \operatorname{Subst}\left[\int \left(-\frac{a}{b}+\frac{x}{b}\right)^m \operatorname{Erfc}[x]^2 \, dx,\ x,\ a+b\,x\right]$$

- **Program code:**

```
Int[x_^m_.*Erfc[a_+b_.*x_]^2,x_Symbol] :=
  Dist[1/b,Subst[Int[(-a/b+x/b)^m*Erfc[x]^2,x],x,a+b*x]] /;
FreeQ[{a,b},x] && IntegerQ[m] && m>0
```

$$\int \frac{x^m \operatorname{Erfc}[b x]}{e^{b^2 x^2}} \, dx$$

- **Derivation: Integration by parts special case**
- **Rule:**

$$\int \frac{x \operatorname{Erfc}[b x]}{e^{b^2 x^2}} \, dx \longrightarrow -\frac{\operatorname{Erfc}[b x]}{2 b^2 e^{b^2 x^2}} - \frac{1}{b \sqrt{\pi}} \int \frac{1}{e^{2 b^2 x^2}} \, dx$$

- **Program code:**

```
Int[x_*E^(c_.*x_^2)*Erfc[b_.*x_],x_Symbol] :=
  -E^(-b^2*x^2)*Erfc[b*x]/(2*b^2) -
  Dist[1/(b*Sqrt[Pi]),Int[E^(-2*b^2*x^2),x]] /;
FreeQ[{b,c},x] && ZeroQ[c+b^2]
```

- **Derivation: Integration by parts**
- **Rule: If** $m \in \mathbb{Z} \wedge m > 1$**, then**

$$\int \frac{x^m \operatorname{Erfc}[b x]}{e^{b^2 x^2}} \, dx \longrightarrow -\frac{x^{m-1} \operatorname{Erfc}[b x]}{2 b^2 e^{b^2 x^2}} - \frac{1}{b \sqrt{\pi}} \int \frac{x^{m-1}}{e^{2 b^2 x^2}} \, dx + \frac{m-1}{2 b^2} \int \frac{x^{m-2} \operatorname{Erfc}[b x]}{e^{b^2 x^2}} \, dx$$

- **Program code:**

```
Int[x_^m_*E^(c_.*x_^2)*Erfc[b_.*x_],x_Symbol] :=
  -x^(m-1)*E^(-b^2*x^2)*Erfc[b*x]/(2*b^2) -
  Dist[1/(b*Sqrt[Pi]),Int[x^(m-1)*E^(-2*b^2*x^2),x]] +
  Dist[(m-1)/(2*b^2),Int[x^(m-2)*E^(-b^2*x^2)*Erfc[b*x],x]] /;
FreeQ[{b,c},x] && ZeroQ[c+b^2] && IntegerQ[m] && m>1
```

- **Derivation: Inverted integration by parts**
- **Rule: If** $\frac{m}{2} \in \mathbb{Z} \wedge m < -1$**, then**

$$\int \frac{x^m \operatorname{Erfc}[b x]}{e^{b^2 x^2}} \, dx \longrightarrow \frac{x^{m+1} \operatorname{Erfc}[b x]}{e^{b^2 x^2} (m+1)} + \frac{2 b}{\sqrt{\pi} (m+1)} \int \frac{x^{m+1}}{e^{2 b^2 x^2}} \, dx + \frac{2 b^2}{m+1} \int \frac{x^{m+2} \operatorname{Erfc}[b x]}{e^{b^2 x^2}} \, dx$$

- **Program code:**

```
Int[x_^m_*E^(c_.*x_^2)*Erfc[b_.*x_],x_Symbol] :=
  x^(m+1)*E^(-b^2*x^2)*Erfc[b*x]/(m+1) +
  Dist[2*b/(Sqrt[Pi]*(m+1)),Int[x^(m+1)*E^(-2*b^2*x^2),x]] +
  Dist[2*b^2/(m+1),Int[x^(m+2)*E^(-b^2*x^2)*Erfc[b*x],x]] /;
FreeQ[{b,c},x] && ZeroQ[c+b^2] && EvenQ[m] && m<-1
```

$$\int \text{Erfi}[a+b\,x]^n \, dx$$

- **Derivation: Integration by parts**
- **Rule:**

$$\int \text{Erfi}[a+b\,x]\,dx \longrightarrow \frac{(a+b\,x)\,\text{Erfi}[a+b\,x]}{b} - \frac{e^{(a+b\,x)^2}}{b\,\sqrt{\pi}}$$

- **Program code:**

```
Int[Erfi[a_.+b_.*x_],x_Symbol] :=
  (a+b*x)*Erfi[a+b*x]/b - E^(a+b*x)^2/(b*Sqrt[Pi]) /;
FreeQ[{a,b},x]
```

- **Derivation: Integration by parts**
- **Rule:**

$$\int \text{Erfi}[a+b\,x]^2\,dx \longrightarrow \frac{(a+b\,x)\,\text{Erfi}[a+b\,x]^2}{b} - \frac{4}{\sqrt{\pi}}\int (a+b\,x)\,e^{(a+b\,x)^2}\,\text{Erfi}[a+b\,x]\,dx$$

- **Program code:**

```
Int[Erfi[a_.+b_.*x_]^2,x_Symbol] :=
  (a+b*x)*Erfi[a+b*x]^2/b -
  Dist[4/Sqrt[Pi],Int[(a+b*x)*E^(a+b*x)^2*Erfi[a+b*x],x]] /;
FreeQ[{a,b},x]
```

$\int x^m \operatorname{Erfi}[a + b x]^n \, dx$

- **Derivation: Integration by parts**
- **Rule: If** $m + 1 \neq 0$**, then**

$$\int x^m \operatorname{Erfi}[a + b x] \, dx \longrightarrow \frac{x^{m+1} \operatorname{Erfi}[a + b x]}{m + 1} - \frac{2 b}{\sqrt{\pi} \, (m + 1)} \int x^{m+1} e^{(a+b x)^2} \, dx$$

- **Program code:**

```
Int[x_^m_.*Erfi[a_.+b_.*x_],x_Symbol] :=
  x^(m+1)*Erfi[a+b*x]/(m+1) -
  Dist[2*b/(Sqrt[Pi]*(m+1)),Int[x^(m+1)*E^(a+b*x)^2,x]] /;
FreeQ[{a,b,m},x] && NonzeroQ[m+1]
```

- **Derivation: Integration by parts**
- **Rule: If** $m \in \mathbb{Z} \bigwedge m + 1 \neq 0 \bigwedge \left(m > 0 \bigvee \frac{m-1}{2} \in \mathbb{Z}\right)$**, then**

$$\int x^m \operatorname{Erfi}[b x]^2 \, dx \longrightarrow \frac{x^{m+1} \operatorname{Erfi}[b x]^2}{m + 1} - \frac{4 b}{\sqrt{\pi} \, (m + 1)} \int x^{m+1} e^{b^2 x^2} \operatorname{Erfi}[b x] \, dx$$

- **Program code:**

```
Int[x_^m_.*Erfi[b_.*x_]^2,x_Symbol] :=
  x^(m+1)*Erfi[b*x]^2/(m+1) -
  Dist[4*b/(Sqrt[Pi]*(m+1)),Int[x^(m+1)*E^(b^2*x^2)*Erfi[b*x],x]] /;
FreeQ[b,x] && IntegerQ[m] && m+1!=0 && (m>0 || OddQ[m])
```

- **Derivation: Integration by substitution**
- **Basis:** $x^m f[a + b x] = \frac{1}{b} \left(-\frac{a}{b} + \frac{a+b x}{b}\right)^m f[a + b x] \, \partial_x (a + b x)$
- **Rule: If** $m \in \mathbb{Z} \bigwedge m > 0$**, then**

$$\int x^m \operatorname{Erfi}[a + b x]^2 \, dx \longrightarrow \frac{1}{b} \operatorname{Subst}\left[\int \left(-\frac{a}{b} + \frac{x}{b}\right)^m \operatorname{Erfi}[x]^2 \, dx, x, a + b x\right]$$

- **Program code:**

```
Int[x_^m_.*Erfi[a_+b_.*x_]^2,x_Symbol] :=
  Dist[1/b,Subst[Int[(-a/b+x/b)^m*Erfi[x]^2,x],x,a+b*x]] /;
FreeQ[{a,b},x] && IntegerQ[m] && m>0
```

$$\int \frac{x^m \, \text{Erfi}[b\,x]}{e^{b^2 x^2}} \, dx$$

- **Derivation: Integration by parts special case**
- **Rule:**

$$\int x\, e^{b^2 x^2}\, \text{Erfi}[b\,x]\, dx \longrightarrow \frac{e^{b^2 x^2}\, \text{Erfi}[b\,x]}{2\,b^2} - \frac{1}{b\,\sqrt{\pi}} \int e^{2\,b^2 x^2}\, dx$$

- **Program code:**

```
Int[x_*E^(c_.*x_^2)*Erfi[b_.*x_],x_Symbol] :=
  E^(b^2*x^2)*Erfi[b*x]/(2*b^2) -
  Dist[1/(b*Sqrt[Pi]),Int[E^(2*b^2*x^2),x]] /;
FreeQ[{b,c},x] && ZeroQ[c-b^2]
```

- **Derivation: Integration by parts**
- **Rule: If** $m \in \mathbb{Z} \land m > 1$**, then**

$$\int x^m\, e^{b^2 x^2}\, \text{Erfi}[b\,x]\, dx \longrightarrow \frac{x^{m-1}\, e^{b^2 x^2}\, \text{Erfi}[b\,x]}{2\,b^2} - \frac{1}{b\,\sqrt{\pi}} \int x^{m-1}\, e^{2\,b^2 x^2}\, dx - \frac{m-1}{2\,b^2} \int x^{m-2}\, e^{b^2 x^2}\, \text{Erfi}[b\,x]\, dx$$

- **Program code:**

```
Int[x_^m_*E^(c_.*x_^2)*Erfi[b_.*x_],x_Symbol] :=
  x^(m-1)*E^(b^2*x^2)*Erfi[b*x]/(2*b^2) -
  Dist[1/(b*Sqrt[Pi]),Int[x^(m-1)*E^(2*b^2*x^2),x]] -
  Dist[(m-1)/(2*b^2),Int[x^(m-2)*E^(b^2*x^2)*Erfi[b*x],x]] /;
FreeQ[{b,c},x] && ZeroQ[c-b^2] && IntegerQ[m] && m>1
```

- **Derivation: Inverted integration by parts**
- **Rule: If** $\frac{m}{2} \in \mathbb{Z} \land m < -1$**, then**

$$\int x^m\, e^{b^2 x^2}\, \text{Erfi}[b\,x]\, dx \longrightarrow \frac{x^{m+1}\, e^{b^2 x^2}\, \text{Erfi}[b\,x]}{m+1} - \frac{2\,b}{\sqrt{\pi}\,(m+1)} \int x^{m+1}\, e^{2\,b^2 x^2}\, dx - \frac{2\,b^2}{m+1} \int x^{m+2}\, e^{b^2 x^2}\, \text{Erfi}[b\,x]\, dx$$

- **Program code:**

```
Int[x_^m_*E^(c_.*x_^2)*Erfi[b_.*x_],x_Symbol] :=
  x^(m+1)*E^(b^2*x^2)*Erfi[b*x]/(m+1) -
  Dist[2*b/(Sqrt[Pi]*(m+1)),Int[x^(m+1)*E^(2*b^2*x^2),x]] -
  Dist[2*b^2/(m+1),Int[x^(m+2)*E^(b^2*x^2)*Erfi[b*x],x]] /;
FreeQ[{b,c},x] && ZeroQ[c-b^2] && EvenQ[m] && m<-1
```