$$\int \frac{1}{a+b\,x^n}\,dx$$

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{1}{a+b\,z^n} = \frac{1}{a} - \frac{b}{a\,(b+a\,z^{-n})}$
- **Rule: If** $n \in \mathbb{F} \land n < 0$**, then**

$$\int \frac{1}{a+b\,x^n}\,dx \longrightarrow \frac{x}{a} - \frac{b}{a}\int \frac{1}{b+a\,x^{-n}}\,dx$$

- **Program code:**

```
Int[1/(a_+b_.*x_^n_),x_Symbol] :=
  x/a - Dist[b/a,Int[1/(b+a*x^(-n)),x]] /;
FreeQ[{a,b},x] && FractionQ[n] && n<0
```

$$\int \frac{1}{\sqrt{a + b\, x^n}}\, dx$$

- **Reference: CRC 278**
- **Derivation: Primitive rule**
- **Basis:** $\text{ArcSinh}'[z] = \frac{1}{\sqrt{1+z^2}}$
- **Rule: If** $a > 0 \land b > 0$**, then**

$$\int \frac{1}{\sqrt{a + b\, x^2}}\, dx \longrightarrow \text{ArcSinh}\left[\frac{\sqrt{b}\, x}{\sqrt{a}}\right]$$

- **Program code:**

```
Int[1/Sqrt[a_+b_.*x_^2],x_Symbol] :=
  ArcSinh[Rt[b,2]*x/Sqrt[a]]/Rt[b,2] /;
FreeQ[{a,b},x] && PositiveQ[a] && PosQ[b]
```

- **Reference: G&R 2.271.4b, CRC 279, A&S 3.3.44**
- **Derivation: Primitive rule**
- **Basis:** $\text{ArcSin}'[z] = \frac{1}{\sqrt{1-z^2}}$
- **Rule: If** $a > 0 \land \neg (b > 0)$**, then**

$$\int \frac{1}{\sqrt{a + b\, x^2}}\, dx \longrightarrow \frac{1}{\sqrt{-b}}\, \text{ArcSin}\left[\frac{\sqrt{-b}\, x}{\sqrt{a}}\right]$$

- **Program code:**

```
Int[1/Sqrt[a_+b_.*x_^2],x_Symbol] :=
  ArcSin[Rt[-b,2]*x/Sqrt[a]]/Rt[-b,2] /;
FreeQ[{a,b},x] && PositiveQ[a] && NegQ[b]
```

- **Reference: CRC 278'**
- **Rule: If $\neg (a > 0) \land b > 0$, then**

$$\int \frac{1}{\sqrt{a + b x^2}} \, dx \longrightarrow \frac{1}{\sqrt{b}} \text{ArcTanh}\left[\frac{\sqrt{b} \, x}{\sqrt{a + b x^2}}\right]$$

- **Program code:**

```
Int[1/Sqrt[a_+b_.*x_^2],x_Symbol] :=
  ArcTanh[Rt[b,2]*x/Sqrt[a+b*x^2]]/Rt[b,2] /;
FreeQ[{a,b},x] && Not[PositiveQ[a]] && PosQ[b]
```

- **Reference: CRC 279'**
- **Rule: If $\neg (a > 0) \land \neg (b > 0)$, then**

$$\int \frac{1}{\sqrt{a + b x^2}} \, dx \longrightarrow \frac{1}{\sqrt{-b}} \text{ArcTan}\left[\frac{\sqrt{-b} \, x}{\sqrt{a + b x^2}}\right]$$

- **Program code:**

```
Int[1/Sqrt[a_+b_.*x_^2],x_Symbol] :=
  ArcTan[Rt[-b,2]*x/Sqrt[a+b*x^2]]/Rt[-b,2] /;
FreeQ[{a,b},x] && Not[PositiveQ[a]] && NegQ[b]
```

- **Rule: If $a > 0$, then**

$$\int \frac{1}{\sqrt{a + b x^4}} \, dx \longrightarrow \frac{1}{\sqrt{a} \left(-\frac{b}{a}\right)^{1/4}} \text{EllipticF}\left[\text{ArcSin}\left[\left(-\frac{b}{a}\right)^{1/4} x\right], -1\right]$$

- **Program code:**

```
Int[1/Sqrt[a_+b_.*x_^4],x_Symbol] :=
  EllipticF[ArcSin[Rt[-b/a,4]*x],-1]/(Sqrt[a]*Rt[-b/a,4]) /;
FreeQ[{a,b},x] && PositiveQ[a]
```

- **Basis:** $\partial_x \frac{\sqrt{a}\sqrt{\frac{a+b x^4}{a}}}{\sqrt{a+b x^4}} = 0$

- **Rule:** If $\neg (a > 0)$, then

$$\int \frac{1}{\sqrt{a+b x^4}} \, dx \longrightarrow \frac{\sqrt{\frac{a+b x^4}{a}}}{\left(-\frac{b}{a}\right)^{1/4} \sqrt{a+b x^4}} \text{EllipticF}\left[\text{ArcSin}\left[\left(-\frac{b}{a}\right)^{1/4} x\right], -1\right]$$

- **Program code:**

```
Int[1/Sqrt[a_+b_.*x_^4],x_Symbol] :=
  Sqrt[(a+b*x^4)/a]/(Rt[-b/a,4]*Sqrt[a+b*x^4])*EllipticF[ArcSin[Rt[-b/a,4]*x],-1] /;
FreeQ[{a,b},x] && Not[PositiveQ[a]]
```

$$\int \frac{x^m}{\sqrt{a+b\,x^n}}\,dx$$

- **Rule: If** $a>0$, **then**

$$\int \frac{x^2}{\sqrt{a+b\,x^4}}\,dx \longrightarrow \frac{1}{\sqrt{a}\,\left(-\frac{b}{a}\right)^{3/4}}\,\text{EllipticE}\left[\text{ArcSin}\left[\left(-\frac{b}{a}\right)^{1/4}x\right],\,-1\right]-\frac{1}{\sqrt{a}\,\left(-\frac{b}{a}\right)^{3/4}}\,\text{EllipticF}\left[\text{ArcSin}\left[\left(-\frac{b}{a}\right)^{1/4}x\right],\,-1\right]$$

- **Program code:**

```
Int[x_^2/Sqrt[a_+b_.*x_^4],x_Symbol] :=
  1/(Sqrt[a]*Rt[-b/a,4]^3)*EllipticE[ArcSin[Rt[-b/a,4]*x],-1] -
  1/(Sqrt[a]*Rt[-b/a,4]^3)*EllipticF[ArcSin[Rt[-b/a,4]*x],-1] /;
FreeQ[{a,b},x] && PositiveQ[a]
```

- **Basis:** $\partial_x \frac{\sqrt{a}\,\sqrt{\frac{a+b\,x^4}{a}}}{\sqrt{a+b\,x^4}}=0$

- **Rule: If** $\neg\,(a>0)$, **then**

$$\int \frac{x^2}{\sqrt{a+b\,x^4}}\,dx \longrightarrow \frac{\sqrt{\frac{a+b\,x^4}{a}}}{\left(-\frac{b}{a}\right)^{3/4}\sqrt{a+b\,x^4}}\,\text{EllipticE}\left[\text{ArcSin}\left[\left(-\frac{b}{a}\right)^{1/4}x\right],\,-1\right]-\frac{\sqrt{\frac{a+b\,x^4}{a}}}{\left(-\frac{b}{a}\right)^{3/4}\sqrt{a+b\,x^4}}\,\text{EllipticF}\left[\text{ArcSin}\left[\left(-\frac{b}{a}\right)^{1/4}x\right],\,-1\right]$$

- **Program code:**

```
Int[x_^2/Sqrt[a_+b_.*x_^4],x_Symbol] :=
  Sqrt[(a+b*x^4)/a]/(Rt[-b/a,4]^3*Sqrt[a+b*x^4])*EllipticE[ArcSin[Rt[-b/a,4]*x],-1] -
  Sqrt[(a+b*x^4)/a]/(Rt[-b/a,4]^3*Sqrt[a+b*x^4])*EllipticF[ArcSin[Rt[-b/a,4]*x],-1] /;
FreeQ[{a,b},x] && Not[PositiveQ[a]]
```

$$\int (a + b\,x^n)^p \, dx$$

- **Derivation: Integration by substitution**
- **Basis:** $\sqrt{a + \frac{b}{z^2}} = -\frac{\sqrt{a+b\left(\frac{1}{z}\right)^2}}{\left(\frac{1}{z}\right)^2}\,\partial_z \frac{1}{z}$
- **Rule:**

$$\int \sqrt{a + \frac{b}{x^2}}\, dx \longrightarrow -\text{Subst}\left[\int \frac{\sqrt{a+b\,x^2}}{x^2}\, dx,\, x,\, \frac{1}{x}\right]$$

- **Program code:**

```
Int[Sqrt[a_.+b_./x_^2],x_Symbol] :=
  -Subst[Int[Sqrt[a+b*x^2]/x^2,x],x,1/x] /;
FreeQ[{a,b},x]
```

- **Reference: G&R 2.110.2', CRC 88d'**
- **Rule: If** $n\,(p+1)+1=0$**, then**

$$\int (a + b\,x^n)^p\, dx \longrightarrow \frac{x\,(a+b\,x^n)^{p+1}}{a}$$

- **Program code:**

```
Int[(a_+b_.*x_^n_)^p_,x_Symbol] :=
  x*(a+b*x^n)^(p+1)/a /;
FreeQ[{a,b,n,p},x] && ZeroQ[n*(p+1)+1]
```

- **Reference: G&R 2.110.1, CRC 88b**
- **Rule: If** $p \in \mathbb{F} \land p > 0 \land n\,p+1 \neq 0$**, then**

$$\int (a + b\,x^n)^p\, dx \longrightarrow \frac{x\,(a+b\,x^n)^p}{n\,p+1} + \frac{a\,n\,p}{n\,p+1}\int (a+b\,x^n)^{p-1}\, dx$$

- **Program code:**

```
Int[(a_+b_.*x_^n_)^p_,x_Symbol] :=
  x*(a+b*x^n)^p/(n*p+1) +
  Dist[a*n*p/(n*p+1),Int[(a+b*x^n)^(p-1),x]] /;
FreeQ[{a,b,n},x] && FractionQ[p] && p>0 && NonzeroQ[n*p+1]
```

- **Reference: G&R 2.110.2, CRC 88d**
- **Rule: If $p \in \mathbb{F} \wedge p < -1$, then**

$$\int (a + b\,x^n)^p\,dx \longrightarrow -\frac{x\,(a + b\,x^n)^{p+1}}{a\,n\,(p+1)} + \frac{n\,(p+1)+1}{a\,n\,(p+1)} \int (a + b\,x^n)^{p+1}\,dx$$

- **Program code:**

```
Int[(a_+b_.*x_^n_)^p_,x_Symbol] :=
  -x*(a+b*x^n)^(p+1)/(n*(p+1)*a) +
  Dist[(n*(p+1)+1)/(a*n*(p+1)),Int[(a+b*x^n)^(p+1),x]] /;
FreeQ[{a,b,n},x] && FractionQ[p] && p<-1
```

- **Reference: G&R 2.110.6, CRC 88c**
- **Rule: If $p \notin \mathbb{Z}$, then**

$$\int \left(a + \frac{b}{x}\right)^p dx \longrightarrow \frac{x\left(a + \frac{b}{x}\right)^{p+1}}{a} + \frac{b\,p}{a} \int \frac{\left(a + \frac{b}{x}\right)^p}{x}\,dx$$

- **Program code:**

```
Int[(a_+b_./x_)^p_,x_Symbol] :=
  x*(a+b/x)^(p+1)/a +
  Dist[b*p/a,Int[(a+b/x)^p/x,x]] /;
FreeQ[{a,b,p},x] && Not[IntegerQ[p]]
```

- **Derivation: Integration by substitution**
- **Note: Transforms p into an integer.**
- **Rule: If $-1 < p < 0 \bigwedge p + \frac{1}{n} \in \mathbb{Z}$, let $q = \text{Denominator}[p]$, then**

$$\int (a + b\,x^n)^p\,dx \longrightarrow \frac{q\,a^{p+\frac{1}{n}}}{n}\,\text{Subst}\left[\int \frac{x^{\frac{q}{n}-1}}{(1 - b\,x^q)^{p+\frac{1}{n}+1}}\,dx,\ x,\ \frac{x^{n/q}}{(a + b\,x^n)^{1/q}}\right]$$

- **Program code:**

```
Int[(a_+b_.*x_^n_)^p_,x_Symbol] :=
  Module[{q=Denominator[p]},
  Dist[q*a^(p+1/n)/n,
    Subst[Int[x^(q/n-1)/(1-b*x^q)^(p+1/n+1),x],x,x^(n/q)/(a+b*x^n)^(1/q)]]] /;
FreeQ[{a,b},x] && RationalQ[{p,n}] && -1<p<0 && IntegerQ[p+1/n]
```

$$\int (a + b (c x^n)^m)^p \, dx$$

- **Derivation: Integration by substitution**
- **Basis:** $f\left[(c x^n)^{1/n}\right] = \frac{x}{(c x^n)^{1/n}} f\left[(c x^n)^{1/n}\right] \partial_x (c x^n)^{1/n}$
- **Basis:** $\partial_x \frac{x}{(c x^n)^{1/n}} = 0$
- **Rule: If** $m\, n \in \mathbb{Z}$**, then**

$$\int (a + b (c x^n)^m)^p \, dx \longrightarrow \frac{x}{(c x^n)^{1/n}} \text{Subst}\left[\int (a + b x^{m n})^p \, dx, x, (c x^n)^{1/n}\right]$$

- **Program code:**

```
Int[(a_+b_.*(c_.*x_^n_)^m_)^p_.,x_Symbol] :=
  Dist[x/(c*x^n)^(1/n),Subst[Int[(a+b*x^(m*n))^p,x],x,(c*x^n)^(1/n)]] /;
FreeQ[{a,b,c,m,n,p},x] && IntegerQ[m*n]
```

- **Derivation: Integration by substitution**
- **Basis:** $f\left[(c x^n)^{1/n}\right] = \frac{x}{(c x^n)^{1/n}} f\left[(c x^n)^{1/n}\right] \partial_x (c x^n)^{1/n}$
- **Basis:** $\partial_x \frac{x}{(c x^n)^{1/n}} = 0$
- **Note: This previously unknown rule not yet implemented.**
- **Rule:**

$$\int f\left[(c x^n)^{1/n}\right] dx \longrightarrow \frac{x}{(c x^n)^{1/n}} \text{Subst}\left[\int f[x] \, dx, x, (c x^n)^{1/n}\right]$$

$$\int x^m (a + b x^n)^p \, dx$$

- **Derivation: Integration by substitution**
- **Basis: If** $m+1 \neq 0$ **and** $\frac{n}{m+1} \in \mathbb{Z}$**, then** $x^m (a + b x^n)^p = \frac{1}{m+1} \left(a + b \left(x^{m+1}\right)^{\frac{n}{m+1}}\right)^p \partial_x x^{m+1}$
- **Rule: If** $m+1 \neq 0 \bigwedge \frac{n}{m+1} \in \mathbb{Z} \bigwedge \frac{n}{m+1} > 1$**, then**

$$\int x^m (a + b x^n)^p \, dx \longrightarrow \frac{1}{m+1} \operatorname{Subst}\left[\int \left(a + b x^{\frac{n}{m+1}}\right)^p dx, x, x^{m+1}\right]$$

- **Program code:**

```
Int[x_^m_.*(a_.+b_.*x_^n_)^p_.,x_Symbol] :=
  Dist[1/(m+1),Subst[Int[(a+b*x^(n/(m+1)))^p,x],x,x^(m+1)]] /;
FreeQ[{a,b,m,n,p},x] && NonzeroQ[m+1] && IntegerQ[n/(m+1)] && n/(m+1)>1 && Not[IntegersQ[m,n,p]]
```

- **Derivation: Algebraic simplification**
- **Basis: If** $p \in \mathbb{Z}$**, then** $x^m (a + b x^n)^p = x^{m+n p} \left(b + \frac{a}{x^n}\right)^p$
- **Rule: If** $p \in \mathbb{Z} \bigwedge p < 0 \bigwedge n \in \mathbb{F} \bigwedge n < 0$**, then**

$$\int x^m (a + b x^n)^p \, dx \longrightarrow \int x^{m+n p} \left(b + \frac{a}{x^n}\right)^p dx$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  Int[x^(m+n*p)*(b+a/x^n)^p,x] /;
FreeQ[{a,b,m},x] && IntegerQ[p] && p<0 && FractionQ[n] && n<0
```

- **Reference: G&R 2.110.3**
- **Derivation: Integration by parts**
- **Rule: If** $m, n \in \mathbb{Z} \bigwedge p \in \mathbb{F} \bigwedge p > 0 \bigwedge ((n > 0 \bigwedge m < -1) \bigvee 0 < -n \leq m+1)$**, then**

$$\int x^m (a + b x^n)^p \, dx \longrightarrow \frac{x^{m+1} (a + b x^n)^p}{m+1} - \frac{b n p}{m+1} \int x^{m+n} (a + b x^n)^{p-1} dx$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  x^(m+1)*(a+b*x^n)^p/(m+1) -
  Dist[b*n*p/(m+1),Int[x^(m+n)*(a+b*x^n)^(p-1),x]] /;
FreeQ[{a,b},x] && IntegersQ[m,n] && FractionQ[p] && p>0 && (n>0 && m<-1 || 0<-n<=m+1)
```

- **Reference: G&R 2.110.4**
- **Derivation: Integration by parts**
- **Basis:** $x^m (a+b x^n)^p = x^{m-n+1} ((a+b x^n)^p x^{n-1})$
- **Rule: If** $m, n \in \mathbb{Z} \land p \in \mathbb{F} \land p < -1 \land (0 < n \le m \lor m \le n < 0) \land m-n+1 \neq 0$**, then**

$$\int x^m (a+b x^n)^p \, dx \longrightarrow \frac{x^{m-n+1} (a+b x^n)^{p+1}}{b n (p+1)} - \frac{m-n+1}{b n (p+1)} \int x^{m-n} (a+b x^n)^{p+1} \, dx$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  x^(m-n+1)*(a+b*x^n)^(p+1)/(b*n*(p+1)) -
  Dist[(m-n+1)/(b*n*(p+1)),Int[x^(m-n)*(a+b*x^n)^(p+1),x]] /;
FreeQ[{a,b},x] && IntegersQ[m,n] && FractionQ[p] && p<-1 && (0<n<=m || m<=n<0) && NonzeroQ[m-n+1]
```

- **Reference: G&R 2.110.1, CRC 88b**
- **Rule: If** $p \in \mathbb{F} \land p > 0 \land m+n p+1 \neq 0 \land \neg \left(\frac{m+1}{n} \in \mathbb{Z} \land \frac{m+1}{n} > 0\right)$**, then**

$$\int x^m (a+b x^n)^p \, dx \longrightarrow \frac{x^{m+1} (a+b x^n)^p}{m+n p+1} + \frac{n p a}{m+n p+1} \int x^m (a+b x^n)^{p-1} \, dx$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  x^(m+1)*(a+b*x^n)^p/(m+n*p+1) +
  Dist[n*p*a/(m+n*p+1),Int[x^m*(a+b*x^n)^(p-1),x]] /;
FreeQ[{a,b,m,n,p},x] && FractionQ[p] && p>0 && NonzeroQ[m+n*p+1] &&
Not[IntegerQ[(m+1)/n] && (m+1)/n>0]
```

- **Reference: G&R 2.110.2, CRC 88d**
- **Rule: If** $p \in \mathbb{F} \land p < -1 \land m+n (p+1)+1 \neq 0 \land m-n+1 \neq 0$**, then**

$$\int x^m (a+b x^n)^p \, dx \longrightarrow -\frac{x^{m+1} (a+b x^n)^{p+1}}{a n (p+1)} + \frac{m+n (p+1)+1}{a n (p+1)} \int x^m (a+b x^n)^{p+1} \, dx$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  -x^(m+1)*(a+b*x^n)^(p+1)/(a*n*(p+1)) +
  Dist[(m+n*(p+1)+1)/(a*n*(p+1)),Int[x^m*(a+b*x^n)^(p+1),x]] /;
FreeQ[{a,b,m,n},x] && FractionQ[p] && p<-1 && NonzeroQ[m+n*(p+1)+1] && NonzeroQ[m-n+1]
```

- **Reference: G&R 2.110.5, CRC 88a**

- **Rule: If $m + n p + 1 \neq 0 \wedge m - n + 1 \neq 0 \wedge m + 1 \neq 0$, then**

$$\int x^m (a + b x^n)^p \, dx \longrightarrow \frac{x^{m-n+1} (a + b x^n)^{p+1}}{b (m + n p + 1)} - \frac{a (m - n + 1)}{b (m + n p + 1)} \int x^{m-n} (a + b x^n)^p \, dx$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*x_^n_.)^p_,x_Symbol] :=
  x^(m-n+1)*(a+b*x^n)^(p+1)/(b*(m+n*p+1)) -
  Dist[a*(m-n+1)/(b*(m+n*p+1)),Int[x^(m-n)*(a+b*x^n)^p,x]] /;
FreeQ[{a,b,m,n,p},x] && NonzeroQ[m+n*p+1] && NonzeroQ[m-n+1] && NonzeroQ[m+1] &&
Not[IntegersQ[m,n,p]] &&
    (IntegersQ[m,n] && (0<n<=m || m<=n<0) && (Not[RationalQ[p]] || -1<p<0) ||
     IntegerQ[(m+1)/n] && 0<(m+1)/n && Not[FractionQ[n]] ||
     Not[RationalQ[m]] && RationalQ[m-n] ||
     RationalQ[n] && MatchQ[m,u_+q_ /; RationalQ[q] && (0<n<=q || n<0 && q<0)] ||
     MatchQ[m,u_+q_.*n /; RationalQ[q] && q>=1])
```

- **Reference: G&R 2.110.6, CRC 88c**

- **Rule: If $m + 1 \neq 0 \wedge m + n (p + 1) + 1 \neq 0$, then**

$$\int x^m (a + b x^n)^p \, dx \longrightarrow \frac{x^{m+1} (a + b x^n)^{p+1}}{a (m + 1)} - \frac{b (m + n (p + 1) + 1)}{a (m + 1)} \int x^{m+n} (a + b x^n)^p \, dx$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*x_^n_.)^p_,x_Symbol] :=
  x^(m+1)*(a+b*x^n)^(p+1)/(a*(m+1)) -
  Dist[b*(m+n*(p+1)+1)/(a*(m+1)),Int[x^(m+n)*(a+b*x^n)^p,x]] /;
FreeQ[{a,b,m,n,p},x] && NonzeroQ[m+1] && NonzeroQ[m+n*(p+1)+1] && Not[IntegersQ[m,n,p]] &&
    (IntegersQ[m,n] && (n>0 && m<-1 || 0<-n<=m+1) ||
     Not[RationalQ[m]] && RationalQ[m+n] ||
     RationalQ[n] && MatchQ[m,u_+q_ /; RationalQ[q] && (n>0 && q<0 || 0<-n<=q)] ||
     MatchQ[m,u_+q_*n /; RationalQ[q] && q<0])
```

- **Derivation: Integration by substitution**
- **Note: Transforms p into an integer.**
- **Rule: If $-1 < p < 0 \bigwedge p + \frac{m+1}{n} \in \mathbb{Z} \bigwedge \mathrm{gcd}[m+1, n] = 1$, let $q = \mathrm{Denominator}[p]$, then**

$$\int x^m (a + b x^n)^p \, dx \longrightarrow \frac{q\, a^{p+\frac{m+1}{n}}}{n} \mathrm{Subst}\left[\int \frac{x^{\frac{q(m+1)}{n}-1}}{(1 - b x^q)^{p+\frac{m+1}{n}+1}} \, dx, x, \frac{x^{n/q}}{(a + b x^n)^{1/q}}\right]$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  Module[{q=Denominator[p]},
  q*a^(p+(m+1)/n)/n*
    Subst[Int[x^(q*(m+1)/n-1)/(1-b*x^q)^(p+(m+1)/n+1),x],x,x^(n/q)/(a+b*x^n)^(1/q)]] /;
FreeQ[{a,b},x] && RationalQ[{m,n,p}] && -1<p<0 && IntegerQ[p+(m+1)/n] && GCD[m+1,n]==1
```

$$\int (x^m (a + b x^n))^p \, dx$$

- **Rule: If** $m p - n + 1 = 0 \land p + 1 \neq 0$**, then**

$$\int (x^m (a + b x^n))^p \, dx \longrightarrow \frac{(x^m (a + b x^n))^{p+1}}{b n (p + 1) x^{m (p+1)}}$$

- **Program code:**

```
Int[(x_^m_.*(a_+b_.*x_^n_.))^p_,x_Symbol] :=
  (x^m*(a+b*x^n))^(p+1)/(b*n*(p+1)*x^(m*(p+1))) /;
FreeQ[{a,b,m,n,p},x] && ZeroQ[m*p-n+1] && NonzeroQ[p+1]
```

- **Rule: If** $m p + n (p + 1) + 1 = 0 \land p + 1 \neq 0$**, then**

$$\int (x^m (a + b x^n))^p \, dx \longrightarrow -\frac{(x^m (a + b x^n))^{p+1}}{a n (p + 1) x^{m-1}}$$

- **Program code:**

```
Int[(x_^m_.*(a_+b_.*x_^n_.))^p_.,x_Symbol] :=
  -(x^m*(a+b*x^n))^(p+1)/(a*n*(p+1)*x^(m-1)) /;
FreeQ[{a,b,m,n,p},x] && ZeroQ[m*p+n*(p+1)+1] && NonzeroQ[p+1]
```

$$\int x^q\,\left(x^m\,\left(a+b\,x^n\right)\right)^p\,dx$$

- **Rule: If** $q+m\,p-n+1=0\,\wedge\,p+1\neq 0$**, then**

$$\int x^q\,\left(x^m\,\left(a+b\,x^n\right)\right)^p\,dx\,\longrightarrow\,\frac{\left(x^m\,\left(a+b\,x^n\right)\right)^{p+1}}{b\,n\,(p+1)\,x^{m\,(p+1)}}$$

- **Program code:**

```
Int[x_^q_.*(x_^m_.*(a_+b_.*x_^n_.))^p_,x_Symbol] :=
  (x^m*(a+b*x^n))^(p+1)/(b*n*(p+1)*x^(m*(p+1))) /;
FreeQ[{a,b,m,n,p},x] && ZeroQ[q+m*p-n+1] && NonzeroQ[p+1]
```

- **Rule: If** $q+m\,p+n\,(p+1)+1=0\,\wedge\,p+1\neq 0$**, then**

$$\int x^q\,\left(x^m\,\left(a+b\,x^n\right)\right)^p\,dx\,\longrightarrow\,-\frac{\left(x^m\,\left(a+b\,x^n\right)\right)^{p+1}}{a\,n\,(p+1)\,x^{m-1-q}}$$

- **Program code:**

```
Int[x_^q_.*(x_^m_.*(a_+b_.*x_^n_.))^p_.,x_Symbol] :=
  -(x^m*(a+b*x^n))^(p+1)/(a*n*(p+1)*x^(m-1-q)) /;
FreeQ[{a,b,m,n,p,q},x] && ZeroQ[q+m*p+n*(p+1)+1] && NonzeroQ[p+1]
```

$$\int (a+b x^2)^m (c+d x^2)^n \, dx$$

- **Derivation: Algebraic simplification**
- **Basis: If** $b c - a d = 0$ **and** $m \in \mathbb{Z}$**, then** $(a+b x^2)^m = \left(\frac{b}{d}\right)^m (c+d x^2)^m$
- **Rule: If** $b c - a d = 0 \wedge m \in \mathbb{Z}$**, then**

$$\int (a+b x^2)^m (c+d x^2)^n \, dx \longrightarrow \left(\frac{b}{d}\right)^m \int (c+d x^2)^{n+m} \, dx$$

- **Program code:**

```
Int[(a_+b_.*x_^2)^m_.*(c_+d_.*x_^2)^n_,x_Symbol] :=
  Dist[(b/d)^m,Int[(c+d*x^2)^(n+m),x]] /;
FreeQ[{a,b,c,d,n},x] && ZeroQ[b*c-a*d] && IntegerQ[m]
```

- **Reference: CRC 232, A&S 3.3.50'**
- **Rule: If** $\frac{a d - b c}{a} > 0$**, then**

$$\int \frac{1}{(a+b x^2)\sqrt{c+d x^2}} \, dx \longrightarrow \frac{1}{a\sqrt{\frac{a d - b c}{a}}} \operatorname{ArcTanh}\left[\frac{x\sqrt{\frac{a d - b c}{a}}}{\sqrt{c+d x^2}}\right]$$

- **Program code:**

```
Int[1/((a_+b_.*x_^2)*Sqrt[c_+d_.*x_^2]),x_Symbol] :=
  ArcTanh[x*Rt[(a*d-b*c)/a,2]/Sqrt[c+d*x^2]]/(a*Rt[(a*d-b*c)/a,2]) /;
FreeQ[{a,b,c,d},x] && PosQ[(a*d-b*c)/a]
```

- **Reference: CRC 233, A&S 3.3.49**
- **Rule: If** $\neg\left(\frac{a d-b c}{a}>0\right)$**, then**

$$\int \frac{1}{\left(a+b x^2\right) \sqrt{c+d x^2}} \, dx \longrightarrow \frac{1}{a \sqrt{\frac{b c-a d}{a}}} \operatorname{ArcTan}\left[\frac{x \sqrt{\frac{b c-a d}{a}}}{\sqrt{c+d x^2}}\right]$$

- **Program code:**

```
Int[1/((a_+b_.*x_^2)*Sqrt[c_+d_.*x_^2]),x_Symbol] :=
  ArcTan[x*Rt[(b*c-a*d)/a,2]/Sqrt[c+d*x^2]]/(a*Rt[(b*c-a*d)/a,2]) /;
FreeQ[{a,b,c,d},x] && NegQ[(a*d-b*c)/a]
```

- **Rule: If** $a>0 \land c>0$**, then**

$$\int \frac{1}{\sqrt{a+b x^2} \sqrt{c+d x^2}} \, dx \longrightarrow \frac{1}{\sqrt{a} \sqrt{c} \sqrt{-\frac{d}{c}}} \operatorname{EllipticF}\left[\operatorname{ArcSin}\left[\sqrt{-\frac{d}{c}} \, x\right], \frac{b c}{a d}\right]$$

- **Program code:**

```
Int[1/(Sqrt[a_+b_.*x_^2]*Sqrt[c_+d_.*x_^2]),x_Symbol] :=
  1/(Sqrt[a]*Sqrt[c]*Rt[-d/c,2])*EllipticF[ArcSin[Rt[-d/c,2]*x], b*c/(a*d)] /;
FreeQ[{a,b,c,d},x] && PositiveQ[a] && PositiveQ[c] &&
(PosQ[-c*d] && (NegQ[-a*b] || Not[RationalQ[Rt[-a*b,2]]]) || NegQ[-c*d] && NegQ[-a*b] &&
Not[RationalQ[Rt[a*b,2]]])
```

- **Derivation: Piecewise constant extraction**

- **Basis:** $\partial_x \frac{\sqrt{\frac{a+b x^2}{a}}}{\sqrt{a+b x^2}} = 0$

- **Rule: If** $\neg (a > 0 \land c > 0)$**, then**

$$\int \frac{1}{\sqrt{a+b x^2}\,\sqrt{c+d x^2}}\,dx \to \frac{\sqrt{\frac{a+b x^2}{a}}\,\sqrt{\frac{c+d x^2}{c}}}{\sqrt{a+b x^2}\,\sqrt{c+d x^2}}\int \frac{1}{\sqrt{1+\frac{b}{a} x^2}\,\sqrt{1+\frac{d}{c} x^2}}\,dx$$

- **Program code:**

```
Int[1/(Sqrt[a_+b_.*x_^2]*Sqrt[c_+d_.*x_^2]),x_Symbol] :=
  Sqrt[(a+b*x^2)/a]*Sqrt[(c+d*x^2)/c]/(Sqrt[a+b*x^2]*Sqrt[c+d*x^2])*Int[1/(Sqrt[1+b/a*x^2]*Sqrt[1+d/
FreeQ[{a,b,c,d},x] && Not[PositiveQ[a] && PositiveQ[c]] &&
(PosQ[-c*d] && (NegQ[-a*b] || Not[RationalQ[Rt[-a*b,2]]]) || NegQ[-c*d] && NegQ[-a*b] &&
Not[RationalQ[Rt[a*b,2]]])
```

- **Rule: If** $a > 0 \land c > 0$**, then**

$$\int \frac{\sqrt{a+b x^2}}{\sqrt{c+d x^2}}\,dx \to \frac{\sqrt{a}}{\sqrt{c}\,\sqrt{-\frac{d}{c}}}\,\text{EllipticE}\left[\text{ArcSin}\left[\sqrt{-\frac{d}{c}}\,x\right],\,\frac{b c}{a d}\right]$$

- **Program code:**

```
Int[Sqrt[a_+b_.*x_^2]/Sqrt[c_+d_.*x_^2],x_Symbol] :=
  Sqrt[a]/(Sqrt[c]*Rt[-d/c,2])*EllipticE[ArcSin[Rt[-d/c,2]*x], b*c/(a*d)] /;
FreeQ[{a,b,c,d},x] && PositiveQ[a] && PositiveQ[c]
```

- **Derivation: Piecewise constant extraction**

- **Basis:** $\partial_x \frac{\sqrt{a+b x^2}}{\sqrt{\frac{a+b x^2}{a}}} = 0$

- **Rule: If** $\neg (a > 0 \land c > 0)$, **then**

$$\int \frac{\sqrt{a+b x^2}}{\sqrt{c+d x^2}} \, dx \longrightarrow \frac{\sqrt{a+b x^2} \sqrt{\frac{c+d x^2}{c}}}{\sqrt{c+d x^2} \sqrt{\frac{a+b x^2}{a}}} \int \frac{\sqrt{1+\frac{b}{a} x^2}}{\sqrt{1+\frac{d}{c} x^2}} \, dx$$

- **Program code:**

```
Int[Sqrt[a_+b_.*x_^2]/Sqrt[c_+d_.*x_^2],x_Symbol] :=
  Sqrt[a+b*x^2]*Sqrt[(c+d*x^2)/c]/(Sqrt[c+d*x^2]*Sqrt[(a+b*x^2)/a])*Int[Sqrt[1+b/a*x^2]/Sqrt[1+d/c*x
FreeQ[{a,b,c,d},x] && Not[PositiveQ[a] && PositiveQ[c]]
```

- **Rule:**

$$\int \sqrt{a+b x^2} \sqrt{c+d x^2} \, dx \longrightarrow$$

$$\frac{x}{3} \sqrt{a+b x^2} \sqrt{c+d x^2} + \frac{c b+a d}{3 d} \int \frac{\sqrt{c+d x^2}}{\sqrt{a+b x^2}} \, dx - \frac{c (c b-a d)}{3 d} \int \frac{1}{\sqrt{a+b x^2} \sqrt{c+d x^2}} \, dx$$

- **Program code:**

```
Int[Sqrt[a_+b_.*x_^2]*Sqrt[c_+d_.*x_^2],x_Symbol] :=
  x/3*Sqrt[a+b*x^2]*Sqrt[c+d*x^2] +
  Dist[(c*b+a*d)/(3*d),Int[Sqrt[c+d*x^2]/Sqrt[a+b*x^2],x]] -
  Dist[c*(c*b-a*d)/(3*d),Int[1/(Sqrt[a+b*x^2]*Sqrt[c+d*x^2]),x]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[c*b+a*d] && NonzeroQ[c*b-a*d]
```

$$\int \left(a+b\,x^2\right)^m \left(c+d\,x^2\right)^n \left(e+f\,x^2\right)^p\,dx$$

- **Derivation: Algebraic simplification**
- **Basis: If** $b\,c-a\,d=0$ **and** $m\in\mathbb{Z}$**, then** $(a+b\,x^2)^m=\left(\frac{b}{d}\right)^m (c+d\,x^2)^m$
- **Rule: If** $b\,c-a\,d=0\,\wedge\,m\in\mathbb{Z}$**, then**

$$\int \left(a+b\,x^2\right)^m \left(c+d\,x^2\right)^n \left(e+f\,x^2\right)^p\,dx \,\longrightarrow\, \left(\frac{b}{d}\right)^m \int \left(c+d\,x^2\right)^{m+n} \left(e+f\,x^2\right)^p\,dx$$

- **Program code:**

```
Int[(a_+b_.*x_^2)^m_.*(c_+d_.*x_^2)^n_*(e_+f_.*x_^2)^p_,x_Symbol] :=
  Dist[(b/d)^m,Int[(c+d*x^2)^(m+n)*(e+f*x^2)^p,x]] /;
FreeQ[{a,b,c,d,e,f,n,p},x] && ZeroQ[b*c-a*d] && IntegerQ[m]
```

- **Rule: If** $c>0\,\wedge\,e>0$**, then**

$$\int \frac{1}{\left(a+b\,x^2\right)\sqrt{c+d\,x^2}\,\sqrt{e+f\,x^2}}\,dx \,\longrightarrow\, \frac{1}{a\,\sqrt{c}\,\sqrt{e}\,\sqrt{-\frac{d}{c}}}\,\text{EllipticPi}\left[\frac{b\,c}{a\,d},\,\text{ArcSin}\left[\sqrt{-\frac{d}{c}}\,x\right],\,\frac{c\,f}{e\,d}\right]$$

- **Program code:**

```
Int[1/((a_+b_.*x_^2)*Sqrt[c_+d_.*x_^2]*Sqrt[e_+f_.*x_^2]),x_Symbol] :=
  1/(a*Sqrt[c]*Sqrt[e]*Rt[-d/c,2])*EllipticPi[b*c/(a*d), ArcSin[Rt[-d/c,2]*x], c*f/(e*d)] /;
FreeQ[{a,b,c,d,e,f},x] && PositiveQ[c] && PositiveQ[e] &&
(PosQ[-e*f] && (NegQ[-c*d] || Not[RationalQ[Rt[-c*d,2]]]) || NegQ[-e*f] && NegQ[-c*d] &&
Not[RationalQ[Rt[c*d,2]]])
```

- **Derivation: Piecewise constant extraction**
- **Basis:** $\partial_x \frac{\sqrt{\frac{c+d x^2}{c}}}{\sqrt{c+d x^2}} = 0$
- **Rule: If** $\neg (c > 0 \land e > 0)$**, then**

$$\int \frac{1}{\left(a+b x^2\right) \sqrt{c+d x^2} \sqrt{e+f x^2}} \, dx \to \frac{\sqrt{\frac{c+d x^2}{c}} \sqrt{\frac{e+f x^2}{e}}}{\sqrt{c+d x^2} \sqrt{e+f x^2}} \int \frac{1}{\left(a+b x^2\right) \sqrt{1+\frac{d}{c} x^2} \sqrt{1+\frac{f}{e} x^2}} \, dx$$

- **Program code:**

```
Int[1/((a_+b_.*x_^2)*Sqrt[c_+d_.*x_^2]*Sqrt[e_+f_.*x_^2]),x_Symbol] :=
  Sqrt[(c+d*x^2)/c]*Sqrt[(e+f*x^2)/e]/(Sqrt[c+d*x^2]*Sqrt[e+f*x^2])*
    Int[1/((a+b*x^2)*Sqrt[1+d/c*x^2]*Sqrt[1+f/e*x^2]),x] /;
FreeQ[{a,b,c,d,e,f},x] && Not[PositiveQ[c] && PositiveQ[e]] &&
(PosQ[-e*f] && (NegQ[-c*d] || Not[RationalQ[Rt[-c*d,2]]]) || NegQ[-e*f] && NegQ[-c*d] &&
Not[RationalQ[Rt[c*d,2]]])
```

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{\sqrt{e+f z}}{(a+b z)} = \frac{f}{b \sqrt{e+f z}} + \frac{b e-a f}{b (a+b z) \sqrt{e+f z}}$
- **Rule: If** $b e - a f \neq 0$**, then**

$$\int \frac{\sqrt{e+f x^2}}{\left(a+b x^2\right) \sqrt{c+d x^2}} \, dx \to \frac{f}{b} \int \frac{1}{\sqrt{c+d x^2} \sqrt{e+f x^2}} \, dx + \frac{b e-a f}{b} \int \frac{1}{\left(a+b x^2\right) \sqrt{c+d x^2} \sqrt{e+f x^2}} \, dx$$

- **Program code:**

```
Int[Sqrt[e_+f_.*x_^2]/((a_+b_.*x_^2)*Sqrt[c_+d_.*x_^2]),x_Symbol] :=
  Dist[f/b,Int[1/(Sqrt[c+d*x^2]*Sqrt[e+f*x^2]),x]] +
  Dist[(b*e-a*f)/b,Int[1/((a+b*x^2)*Sqrt[c+d*x^2]*Sqrt[e+f*x^2]),x]] /;
FreeQ[{a,b,c,d,e,f},x] && NonzeroQ[b*e-a*f]
```

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{\sqrt{c+d x^2}\,\sqrt{e+f x^2}}{a+b x^2}=\frac{d\sqrt{e+f x^2}}{b\sqrt{c+d x^2}}+\frac{(b c-a d)\sqrt{e+f x^2}}{b\,(a+b x^2)\sqrt{c+d x^2}}$
- **Rule: If** $b c-a d\neq 0$**, then**

$$\int\frac{\sqrt{c+d x^2}\,\sqrt{e+f x^2}}{a+b x^2}\,dx\,\longrightarrow\,\frac{d}{b}\int\frac{\sqrt{e+f x^2}}{\sqrt{c+d x^2}}\,dx+\frac{b c-a d}{b}\int\frac{\sqrt{e+f x^2}}{(a+b x^2)\sqrt{c+d x^2}}\,dx$$

- **Program code:**

```
Int[Sqrt[c_+d_.*x_^2]*Sqrt[e_+f_.*x_^2]/(a_+b_.*x_^2),x_Symbol] :=
  Dist[d/b,Int[Sqrt[e+f*x^2]/Sqrt[c+d*x^2],x]] +
  Dist[(b*c-a*d)/b,Int[Sqrt[e+f*x^2]/((a+b*x^2)*Sqrt[c+d*x^2]),x]] /;
FreeQ[{a,b,c,d,e,f},x] && NonzeroQ[b*c-a*d]
```

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{e+f z}{\sqrt{a+b z}}=\frac{f\sqrt{a+b z}}{b}+\frac{b e-a f}{b\sqrt{a+b z}}$
- **Rule: If** $b e-a f\neq 0$**, then**

$$\int\frac{e+f x^2}{\sqrt{a+b x^2}\,\sqrt{c+d x^2}}\,dx\,\longrightarrow\,\frac{f}{b}\int\frac{\sqrt{a+b x^2}}{\sqrt{c+d x^2}}\,dx+\frac{b e-a f}{b}\int\frac{1}{\sqrt{a+b x^2}\,\sqrt{c+d x^2}}\,dx$$

- **Program code:**

```
Int[(e_.+f_.*x_^2)/(Sqrt[a_+b_.*x_^2]*Sqrt[c_+d_.*x_^2]),x_Symbol] :=
  Dist[f/b,Int[Sqrt[a+b*x^2]/Sqrt[c+d*x^2],x]] +
  Dist[(b*e-a*f)/b,Int[1/(Sqrt[a+b*x^2]*Sqrt[c+d*x^2]),x]] /;
FreeQ[{a,b,c,d,e,f},x] && NonzeroQ[b*e-a*f] &&
(PosQ[-c*d] && (NegQ[-a*b] || Not[RationalQ[Rt[-a*b,2]]]) || NegQ[-c*d] && NegQ[-a*b] &&
Not[RationalQ[Rt[a*b,2]]])
```

- **Rule:**

$$\int \frac{x^2 \sqrt{a+b x^2}}{\sqrt{c+d x^2}} \, dx \longrightarrow \frac{x \sqrt{a+b x^2} \sqrt{c+d x^2}}{3 d} - \frac{1}{3 d} \int \frac{a c + (2 b c - a d) x^2}{\sqrt{a+b x^2} \sqrt{c+d x^2}} \, dx$$

- **Program code:**

```
Int[x_^2*Sqrt[a_+b_.*x_^2]/Sqrt[c_+d_.*x_^2],x_Symbol] :=
  x*Sqrt[a+b*x^2]*Sqrt[c+d*x^2]/(3*d) -
  Dist[1/(3*d),Int[(a*c+(2*b*c-a*d)*x^2)/(Sqrt[a+b*x^2]*Sqrt[c+d*x^2]),x]] /;
FreeQ[{a,b,c,d},x]
```

$$\int (a+b\,x^n)^p\,(c+d\,x^n)^q\,dx$$

- **Derivation: Algebraic simplification**
- **Basis: If** $a\,d-b\,c=0$**, then** $\frac{a+b\,z}{c+d\,z}=\frac{b}{d}$
- **Rule: If** $a\,d-b\,c=0\,\land\,p>0\,\land\,q<-1$**, then**

$$\int (a+b\,x^n)^p\,(c+d\,x^n)^q\,dx\;\longrightarrow\;\frac{b}{d}\int (a+b\,x^n)^{p-1}\,(c+d\,x^n)^{q+1}\,dx$$

- **Program code:**

```
Int[(a_.+b_.*x_^n_)^p_*(c_.+d_.*x_^n_)^q_.,x_Symbol] :=
  Dist[b/d,Int[(a+b*x^n)^(p-1)*(c+d*x^n)^(q+1),x]] /;
FreeQ[{a,b,c,d,n},x] && ZeroQ[a*d-b*c] && RationalQ[{p,q}] && p>0 && q<=-1
```

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{a+b\,z}{c+d\,z}=\frac{b}{d}+\frac{a\,d-b\,c}{d\,(c+d\,z)}$
- **Rule: If** $a\,d-b\,c\neq 0\,\land\,p>0\,\land\,q<-1\,\land\,n\in\mathbb{Z}\,\land\,n>0$**, then**

$$\int (a+b\,x^n)^p\,(c+d\,x^n)^q\,dx\;\longrightarrow\;\frac{a\,d-b\,c}{d}\int (a+b\,x^n)^{p-1}\,(c+d\,x^n)^q\,dx+\frac{b}{d}\int (a+b\,x^n)^{p-1}\,(c+d\,x^n)^{q+1}\,dx$$

- **Program code:**

```
Int[(a_.+b_.*x_^n_)^p_*(c_.+d_.*x_^n_)^q_.,x_Symbol] :=
  Dist[(a*d-b*c)/d,Int[(a+b*x^n)^(p-1)*(c+d*x^n)^q,x]] +
  Dist[b/d,Int[(a+b*x^n)^(p-1)*(c+d*x^n)^(q+1),x]] /;
FreeQ[{a,b,c,d,n},x] && NonzeroQ[a*d-b*c] && RationalQ[{p,q}] && p>0 && q<=-1 &&
IntegerQ[n] && n>0
```

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{1}{(a+b z)\ (c+d z)} = \frac{b}{(b c-a d)\ (a+b z)} - \frac{d}{(b c-a d)\ (c+d z)}$
- **Rule: If** $a d - b c \neq 0 \land p < -1 \land q < -1 \land n \in \mathbb{Z} \land n > 0$**, then**

$$\int (a + b x^n)^p (c + d x^n)^q \, dx \longrightarrow$$
$$\frac{b}{b c - a d} \int (a + b x^n)^p (c + d x^n)^{q+1} \, dx - \frac{d}{b c - a d} \int (a + b x^n)^{p+1} (c + d x^n)^q \, dx$$

- **Program code:**

```
Int[(a_.+b_.*x_^n_)^p_.*(c_.+d_.*x_^n_)^q_.,x_Symbol] :=
  Dist[b/(b*c-a*d),Int[(a+b*x^n)^p*(c+d*x^n)^(q+1),x]] -
  Dist[d/(b*c-a*d),Int[(a+b*x^n)^(p+1)*(c+d*x^n)^q,x]] /;
FreeQ[{a,b,c,d,n},x] && NonzeroQ[b*c-a*d] && RationalQ[{p,q}] && p<-1 && q<=-1 &&
IntegerQ[n] && n>0
```

$$\int \frac{x^m \left(a + b x^n\right)}{c + d x^p} \, dx$$

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{a+b x^n}{x \left(c+d x^p\right)} = \frac{a}{c x} + \frac{x^{n-1} \left(b c-a d x^{p-n}\right)}{c \left(c+d x^p\right)}$
- **Rule: If $n, p \in \mathbb{F} \wedge 0 < n \leq p$, then**

$$\int \frac{a + b x^n}{x \left(c + d x^p\right)} \, dx \longrightarrow \frac{a \operatorname{Log}[x]}{c} + \frac{1}{c} \int \frac{x^{n-1} \left(b c - a d x^{p-n}\right)}{c + d x^p} \, dx$$

- **Program code:**

```
Int[(a_.+b_.*x_^n_.)/(x_*(c_+d_.*x_^p_.)),x_Symbol] :=
  a*Log[x]/c +
  Dist[1/c,Int[x^(n-1)*(b*c-a*d*x^(p-n))/(c+d*x^p),x]] /;
FreeQ[{a,b,c,d},x] && FractionQ[{n,p}] && 0<n<=p
```

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{a+b x^n}{x \left(c+d x^p\right)} = \frac{a}{c x} + \frac{x^{p-1} \left(-a d+b c x^{n-p}\right)}{c \left(c+d x^p\right)}$
- **Rule: If $n, p \in \mathbb{F} \wedge 0 < p < n$, then**

$$\int \frac{a + b x^n}{x \left(c + d x^p\right)} \, dx \longrightarrow \frac{a \operatorname{Log}[x]}{c} + \frac{1}{c} \int \frac{x^{p-1} \left(-a d + b c x^{n-p}\right)}{c + d x^p} \, dx$$

- **Program code:**

```
Int[(a_.+b_.*x_^n_.)/(x_*(c_+d_.*x_^p_.)),x_Symbol] :=
  a*Log[x]/c +
  Dist[1/c,Int[x^(p-1)*(-a*d+b*c*x^(n-p))/(c+d*x^p),x]] /;
FreeQ[{a,b,c,d},x] && FractionQ[{n,p}] && 0<p<n
```

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{x^m (a+b x^n)}{c+d x^p} = \frac{a x^m}{c} + \frac{x^{m+n} (b c-a d x^{p-n})}{c (c+d x^p)}$
- **Rule: If** $m, n, p \in \mathbb{F} \land m < -1 \land 0 < n \leq p$**, then**

$$\int \frac{x^m (a+b x^n)}{c+d x^p} \, dx \longrightarrow \frac{a x^{m+1}}{c (m+1)} + \frac{1}{c} \int \frac{x^{m+n} (b c-a d x^{p-n})}{c+d x^p} \, dx$$

- **Program code:**

```
Int[x_^m_*(a_.+b_.*x_^n_.)/(c_+d_.*x_^p_.),x_Symbol] :=
  a*x^(m+1)/(c*(m+1)) +
  Dist[1/c,Int[x^(m+n)*(b*c-a*d*x^(p-n))/(c+d*x^p),x]] /;
FreeQ[{a,b,c,d},x] && FractionQ[{m,n,p}] && m<-1 && 0<n<=p
```

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{x^m (a+b x^n)}{c+d x^p} = \frac{a x^m}{c} + \frac{x^{m+p} (-a d+b c x^{n-p})}{c (c+d x^p)}$
- **Rule: If** $m, n, p \in \mathbb{F} \land m < -1 \land 0 < p < n$**, then**

$$\int \frac{x^m (a+b x^n)}{c+d x^p} \, dx \longrightarrow \frac{a x^{m+1}}{c (m+1)} + \frac{1}{c} \int \frac{x^{m+p} (-a d+b c x^{n-p})}{c+d x^p} \, dx$$

- **Program code:**

```
Int[x_^m_*(a_.+b_.*x_^n_.)/(c_+d_.*x_^p_.),x_Symbol] :=
  a*x^(m+1)/(c*(m+1)) +
  Dist[1/c,Int[x^(m+p)*(-a*d+b*c*x^(n-p))/(c+d*x^p),x]] /;
FreeQ[{a,b,c,d},x] && FractionQ[{m,n,p}] && m<-1 && 0<p<n
```

$$\int \frac{1}{\sqrt{c\,x^2\,(a+b\,x^n)}}\,dx$$

- **Derivation: Algebraic simplification**

- **Note: If $\frac{b^2 c}{a} < 0$, antiderivative can be expressed more simply as the arcsine or hyperbolic arcsine of a linear term**

- **Rule: If $\frac{b^2 c}{a} < 0$, then**

$$\int \frac{1}{\sqrt{c\,x^2\left(a+\frac{b}{x}\right)}}\,dx \longrightarrow \int \frac{1}{\sqrt{b\,c\,x+a\,c\,x^2}}\,dx$$

- **Program code:**

```
Int[1/Sqrt[c_.*x_^2*(a_.+b_./x_)],x_Symbol] :=
  Int[1/Sqrt[b*c*x+a*c*x^2],x] /;
FreeQ[{a,b,c},x] && NegativeQ[b^2*c/a]
```

- **Rule: If $a\,c > 0$, then**

$$\int \frac{1}{\sqrt{c\,x^2\,(a+b\,x^n)}}\,dx \longrightarrow -\frac{2}{n\sqrt{a\,c}}\,\text{ArcTanh}\left[\frac{\sqrt{a\,c}\;x}{\sqrt{c\,x^2\,(a+b\,x^n)}}\right]$$

- **Program code:**

```
Int[1/Sqrt[c_.*x_^2*(a_.+b_.*x_^n_.)],x_Symbol] :=
  -2/(n*Rt[a*c,2])*ArcTanh[Rt[a*c,2]*x/Sqrt[c*x^2*(a+b*x^n)]] /;
FreeQ[{a,b,c,n},x] && PosQ[a*c]
```

- **Rule: If $\neg\,(a\,c > 0)$, then**

$$\int \frac{1}{\sqrt{c\,x^2\,(a+b\,x^n)}}\,dx \longrightarrow -\frac{2}{n\sqrt{-a\,c}}\,\text{ArcTan}\left[\frac{\sqrt{-a\,c}\;x}{\sqrt{c\,x^2\,(a+b\,x^n)}}\right]$$

- **Program code:**

```
Int[1/Sqrt[c_.*x_^2*(a_.+b_.*x_^n_.)],x_Symbol] :=
  -2/(n*Rt[-a*c,2])*ArcTan[Rt[-a*c,2]*x/Sqrt[c*x^2*(a+b*x^n)]] /;
FreeQ[{a,b,c,n},x] && NegQ[a*c]
```

- **Derivation: Algebraic simplification**
- **Rule: If $m + n = 2$, then**

$$\int \frac{1}{\sqrt{c\,x^m\,(a + b\,x^n)}}\,dx \longrightarrow \int \frac{1}{\sqrt{c\,x^2\,\left(b + a\,x^{m-2}\right)}}\,dx$$

- **Program code:**

```
Int[1/Sqrt[c_.*x_^m_.*(a_.+b_.*x_^n_.)],x_Symbol] :=
  Int[1/Sqrt[c*x^2*(b+a*x^(m-2))],x] /;
FreeQ[{a,b,c,m,n},x] && ZeroQ[m+n-2]
```

- **Derivation: Algebraic simplification**
- **Rule:**

$$\int \frac{1}{\sqrt{c\,\left(a\,x^p + b\,x^2\right)}}\,dx \longrightarrow \int \frac{1}{\sqrt{c\,x^2\,\left(b + a\,x^{p-2}\right)}}\,dx$$

- **Program code:**

```
Int[1/Sqrt[c_.*(a_.*x_^p_+b_.*x_^2)],x_Symbol] :=
  Int[1/Sqrt[c*x^2*(b+a*x^(p-2))],x] /;
FreeQ[{a,b,c,p},x]
```

- **Derivation: Algebraic simplification**
- **Basis:** $x^m\,(a\,x^p + b\,x^{2-m}) = x^2\,(b + a\,x^{m+p-2})$
- **Rule: If $m + n = 2$, then**

$$\int \frac{1}{\sqrt{c\,x^m\,(a\,x^p + b\,x^n)}}\,dx \longrightarrow \int \frac{1}{\sqrt{c\,x^2\,\left(b + a\,x^{m+p-2}\right)}}\,dx$$

- **Program code:**

```
Int[1/Sqrt[c_.*x_^m_.*(a_.*x_^p_.+b_.*x_^n_.)],x_Symbol] :=
  Int[1/Sqrt[c*x^2*(b+a*x^(m+p-2))],x] /;
FreeQ[{a,b,c,m,n,p},x] && ZeroQ[m+n-2]
```

$$\int x^m \left(\frac{a + b x}{c + d x}\right)^n dx$$

- **Program code:**

```
(* Int[(e_.*(a_.+b_.*x_)/(c_.+d_.*x_))^n_,x_Symbol] :=
  Dist[e*(b*c-a*d),Subst[Int[x^n/(b*e-d*x)^2,x],x,e*(a+b*x)/(c+d*x)]] /;
FreeQ[{a,b,c,d,e},x] && FractionQ[n] && NonzeroQ[b*c-a*d] *)
```

- **Program code:**

```
(* Int[x_^m_.*(e_.*(a_.+b_.*x_)/(c_.+d_.*x_))^n_,x_Symbol] :=
  Dist[e*(b*c-a*d),Subst[Int[x^n*(-a*e+c*x)^m/(b*e-d*x)^(m+2),x],x,e*(a+b*x)/(c+d*x)]] /;
FreeQ[{a,b,c,d,e},x] && IntegerQ[m] && FractionQ[n] && NonzeroQ[b*c-a*d] *)
```

- **Program code:**

```
(* Int[(f_+g_.*x_)^m_*(e_.*(a_.+b_.*x_)/(c_.+d_.*x_))^n_,x_Symbol] :=
  Dist[1/g,Subst[Int[x^m*(e*(a-b*f/g+b/g*x)/(c-d*f/g+d/g*x))^n,x],x,f+g*x]] /;
FreeQ[{a,b,c,d,e,f,g},x] && IntegerQ[m] && m<0 && FractionQ[n] && NonzeroQ[b*c-a*d] *)
```

$$\int \sqrt{a\,x + \sqrt{b + a^2\,x^2}}\,dx$$

- **Rule:**

$$\int \sqrt{a\,x + \sqrt{b + a^2\,x^2}}\,dx \longrightarrow \frac{2}{3\,a}\left(2\,a\,x - \sqrt{b + c\,x^2}\right)\sqrt{a\,x + \sqrt{b + c\,x^2}}$$

- **Program code:**

```
Int[Sqrt[a_.*x_+Sqrt[b_+c_.*x_^2]], x_Symbol] :=
  2*(2*a*x-Sqrt[b+c*x^2])*Sqrt[a*x+Sqrt[b+c*x^2]]/(3*a) /;
FreeQ[{a,b,c},x] && ZeroQ[c-a^2]
```

- **Rule:**

$$\int \sqrt{a\,x - \sqrt{b + a^2\,x^2}}\,dx \longrightarrow \frac{2}{3\,a}\left(2\,a\,x + \sqrt{b + c\,x^2}\right)\sqrt{a\,x - \sqrt{b + c\,x^2}}$$

- **Program code:**

```
Int[Sqrt[a_.*x_-Sqrt[b_+c_.*x_^2]], x_Symbol] :=
  2*(2*a*x+Sqrt[b+c*x^2])*Sqrt[a*x-Sqrt[b+c*x^2]]/(3*a) /;
FreeQ[{a,b,c},x] && ZeroQ[c-a^2]
```

$$\int \sqrt{a+\sqrt{a^2+b\,x^2}}\,dx$$

- **Rule:**

$$\int \sqrt{a+\sqrt{a^2+b\,x^2}}\,dx \longrightarrow \frac{2}{3\,b\,x}\left(-a^2+b\,x^2+a\sqrt{a^2+b\,x^2}\right)\sqrt{a+\sqrt{a^2+b\,x^2}}$$

- **Program code:**

```
Int[Sqrt[a_+Sqrt[c_+b_.*x_^2]], x_Symbol] :=
  2*(-a^2+b*x^2+a*Sqrt[a^2+b*x^2])*Sqrt[a+Sqrt[a^2+b*x^2]]/(3*b*x) /;
FreeQ[{a,b,c},x] && ZeroQ[c-a^2]
```

- **Rule:**

$$\int \sqrt{a-\sqrt{a^2+b\,x^2}}\,dx \longrightarrow \frac{2}{3\,b\,x}\left(-a^2+b\,x^2-a\sqrt{a^2+b\,x^2}\right)\sqrt{a-\sqrt{a^2+b\,x^2}}$$

- **Program code:**

```
Int[Sqrt[a_-Sqrt[c_+b_.*x_^2]], x_Symbol] :=
  2*(-a^2+b*x^2-a*Sqrt[a^2+b*x^2])*Sqrt[a-Sqrt[a^2+b*x^2]]/(3*b*x) /;
FreeQ[{a,b,c},x] && ZeroQ[c-a^2]
```

$$\int \frac{u}{v+\sqrt{w}}\, dx$$

- **Derivation: Algebraic simplification**
- **Basis:** $\frac{1}{z+w} = \frac{z-w}{z^2-w^2}$
- **Rule:**

$$\int \frac{u}{a\,x^m+b\,\sqrt{c\,x^n}}\, dx \longrightarrow \int \frac{u\,\left(a\,x^m-b\,\sqrt{c\,x^n}\right)}{a^2\,x^{2\,m}-b^2\,c\,x^n}\, dx$$

- **Program code:**

```
Int[u_./(a_.*x_^m_.+b_.*Sqrt[c_.*x_^n_]),x_Symbol] :=
  Int[u*(a*x^m-b*Sqrt[c*x^n])/(a^2*x^(2*m)-b^2*c*x^n),x] /;
FreeQ[{a,b,c,m,n},x]
```

- **Derivation: Algebraic simplification**
- **Basis: If** $b\,e^2 = d\,f^2$**, then** $\frac{1}{e\,\sqrt{a+b\,z}+f\,\sqrt{c+d\,z}} = \frac{e\,\sqrt{a+b\,z}-f\,\sqrt{c+d\,z}}{a\,e^2-c\,f^2}$
- **Rule: If** $m \in \mathbb{Z} \land m < 0 \land b\,e^2 = d\,f^2$**, then**

$$\int u\,\left(e\,\sqrt{a+b\,x^n}+f\,\sqrt{c+d\,x^n}\right)^m dx \longrightarrow \left(a\,e^2-c\,f^2\right)^m \int \frac{u}{\left(e\,\sqrt{a+b\,x^n}-f\,\sqrt{c+d\,x^n}\right)^m}\, dx$$

- **Program code:**

```
Int[u_.*(e_.*Sqrt[a_.+b_.*x_^n_.]+f_.*Sqrt[c_.+d_.*x_^n_.])^m_,x_Symbol] :=
  Dist[(a*e^2-c*f^2)^m,Int[u*(e*Sqrt[a+b*x^n]-f*Sqrt[c+d*x^n])^(-m),x]] /;
FreeQ[{a,b,c,d,e,f,n},x] && IntegerQ[m] && m<0 && ZeroQ[b*e^2-d*f^2]
```

- **Derivation: Algebraic simplification**
- **Basis: If** $a e^2 = c f^2$**, then** $\frac{1}{e\sqrt{a+b z}+f\sqrt{c+d z}} = \frac{e\sqrt{a+b z}-f\sqrt{c+d z}}{\left(b e^2-d f^2\right) z}$
- **Rule: If** $m \in \mathbb{Z} \bigwedge m < 0 \bigwedge a e^2 = c f^2$**, then**

$$\int u \left(e\sqrt{a+b x^n}+f\sqrt{c+d x^n}\right)^m dx \longrightarrow \left(b e^2-d f^2\right)^m \int \frac{u x^{m n}}{\left(e\sqrt{a+b x^n}-f\sqrt{c+d x^n}\right)^m} dx$$

- **Program code:**

```
Int[u_.*(e_.*Sqrt[a_.+b_.*x_^n_.]+f_.*Sqrt[c_.+d_.*x_^n_.])^m_,x_Symbol] :=
  Dist[(b*e^2-d*f^2)^m,Int[u*x^(m*n)*(e*Sqrt[a+b*x^n]-f*Sqrt[c+d*x^n])^(-m),x]] /;
FreeQ[{a,b,c,d,e,f,n},x] && IntegerQ[m] && m<0 && ZeroQ[a*e^2-c*f^2]
```

- **Derivation: Algebraic simplification**
- **Basis: If** $a^2 = b^2 c$**, then** $\frac{1}{a+b\sqrt{c+d z}} = -\frac{a}{b^2 d z} + \frac{\sqrt{c+d z}}{b d z}$
- **Rule: If** $a^2 = b^2 c$**, then**

$$\int \frac{u}{a+b\sqrt{c+d x^n}} dx \longrightarrow -\frac{a}{b^2 d} \int \frac{u}{x^n} dx + \frac{1}{b d} \int \frac{u\sqrt{c+d x^n}}{x^n} dx$$

- **Program code:**

```
Int[u_./(a_+b_.*Sqrt[c_+d_.*x_^n_]),x_Symbol] :=
  Dist[-a/(b^2*d),Int[u/x^n,x]] +
  Dist[1/(b*d),Int[u*Sqrt[c+d*x^n]/x^n,x]] /;
FreeQ[{a,b,c,d,n},x] && ZeroQ[a^2-b^2*c]
```

- **Derivation: Algebraic simplification**
- **Basis: If** $a^2 = b^2 d$**, then** $\frac{1}{a z+b \sqrt{c+d z^2}} = \frac{-a z+b \sqrt{c+d z^2}}{b^2 c}$
- **Rule: If** $a^2 = b^2 d$**, then**

$$\int \frac{u}{a x^m + b \sqrt{c + d x^{2 m}}} \, dx \longrightarrow -\frac{a}{b^2 c} \int u x^m \, dx + \frac{1}{b c} \int u \sqrt{c + d x^{2 m}} \, dx$$

- **Program code:**

```
Int[u_./(a_.*x_^m_.+b_.*Sqrt[c_+d_.*x_^n_]),x_Symbol] :=
  Dist[-a/(b^2*c),Int[u*x^m,x]] +
  Dist[1/(b*c),Int[u*Sqrt[c+d*x^n],x]] /;
FreeQ[{a,b,c,d,m,n},x] && ZeroQ[n-2*m] && ZeroQ[a^2-b^2*d]
```

- **Derivation: Algebraic simplification**
- **Basis: If** $a^2 = c^2 d \wedge b^2 = c^2 e$**, then** $\frac{1}{a+b z+c \sqrt{d+e z^2}} = \frac{1}{2 a} + \frac{1}{2 b z} - \frac{c \sqrt{d+e z^2}}{2 a b z}$
- **Rule: If** $a^2 = c^2 d \wedge b^2 = c^2 e$**, then**

$$\int \frac{u}{a + b x^m + c \sqrt{d + e x^{2 m}}} \, dx \longrightarrow \frac{1}{2 a} \int u \, dx + \frac{1}{2 b} \int \frac{u}{x^m} \, dx - \frac{c}{2 a b} \int \frac{u \sqrt{d + e x^{2 m}}}{x^m} \, dx$$

- **Program code:**

```
Int[u_./(a_+b_.*x_^m_.+c_.*Sqrt[d_+e_.*x_^n_]),x_Symbol] :=
  Dist[1/(2*a),Int[u,x]] +
  Dist[1/(2*b),Int[u/x^m,x]] -
  Dist[c/(2*a*b),Int[u*Sqrt[d+e*x^n]/x^m,x]] /;
FreeQ[{a,b,c,d,m,n},x] && ZeroQ[n-2*m] && ZeroQ[a^2-c^2*d] && ZeroQ[b^2-c^2*e]
```

- **Derivation: Algebraic simplification**
- **Basis: If** $a^2 = c^2 d \land 2 a b = c^2 e$**, then** $\frac{1}{a+b z+c \sqrt{d+e z}} = \frac{1}{b z} + \frac{a}{b^2 z^2} - \frac{c \sqrt{d+e z}}{b^2 z^2}$
- **Rule: If** $a^2 = c^2 d \land 2 a b = c^2 e$**, then**

$$\int \frac{u}{a + b x^n + c \sqrt{d + e x^n}} \, dx \longrightarrow \frac{1}{b} \int \frac{u}{x^n} \, dx + \frac{a}{b^2} \int \frac{u}{x^{2 n}} \, dx - \frac{c}{b^2} \int \frac{u \sqrt{d + e x^n}}{x^{2 n}} \, dx$$

- **Program code:**

```
Int[u_./(a_+b_.*x_^n_.+c_.*Sqrt[d_+e_.*x_^n_.]),x_Symbol] :=
  Dist[1/b,Int[u/x^n,x]] +
  Dist[a/b^2,Int[u/x^(2*n),x]] -
  Dist[c/b^2,Int[u*Sqrt[d+e*x^n]/x^(2*n),x]] /;
FreeQ[{a,b,c,d,n},x] && ZeroQ[a^2-c^2*d] && ZeroQ[2*a*b-c^2*e]
```

```
(* Int[u_./(e_.*Sqrt[a_.+b_.*x_]+f_.*Sqrt[c_.+d_.*x_]),x_Symbol] :=
  Int[u*(e*Sqrt[a+b*x]-f*Sqrt[c+d*x])/(a*e^2-c*f^2+(b*e^2-d*f^2)*x),x] /;
FreeQ[{a,b,c,d,e,f},x] *)
```

- **Derivation: Algebraic simplification**
- **Basis:** $\frac{1}{a x+b \sqrt{c+d x^2}} = \frac{a x}{-b^2 c+\left(a^2-b^2 d\right) x^2} - \frac{b \sqrt{c+d x^2}}{-b^2 c+\left(a^2-b^2 d\right) x^2}$
- **Rule:**

$$\int \frac{u}{a x + b \sqrt{c + d x^2}} \, dx \longrightarrow a \int \frac{x u}{-b^2 c + \left(a^2 - b^2 d\right) x^2} \, dx - b \int \frac{u \sqrt{c + d x^2}}{-b^2 c + \left(a^2 - b^2 d\right) x^2} \, dx$$

- **Program code:**

```
Int[u_./(a_.*x_+b_.*Sqrt[c_.+d_.*x_^2]),x_Symbol] :=
  Dist[a,Int[x*u/(-b^2*c+(a^2-b^2*d)*x^2),x]] -
  Dist[b,Int[u*Sqrt[c+d*x^2]/(-b^2*c+(a^2-b^2*d)*x^2),x]] /;
FreeQ[{a,b,c,d},x]
```

- **Derivation: Algebraic simplification**
- **Basis:** $\frac{1}{z+w} = \frac{z-w}{z^2-w^2}$
- **Rule:**

$$\int \frac{u}{e\sqrt{(a+b\,x^n)^p} + f\sqrt{(a+b\,x^n)^q}}\,dx \longrightarrow \int \frac{u\left(e\sqrt{(a+b\,x^n)^p} - f\sqrt{(a+b\,x^n)^q}\right)}{e^2\,(a+b\,x^n)^p - f^2\,(a+b\,x^n)^q}\,dx$$

- **Program code:**

```
Int[u_./(e_.*Sqrt[(a_.+b_.*x_^n_.)^p_.]+f_.*Sqrt[(a_.+b_.*x_^n_.)^q_.]),x_Symbol] :=
  Int[u*(e*Sqrt[(a+b*x^n)^p]-f*Sqrt[(a+b*x^n)^q])/(e^2*(a+b*x^n)^p-f^2*(a+b*x^n)^q),x] /;
FreeQ[{a,b,e,f},x] && IntegersQ[n,p,q]
```

```
(* Int[u_./(v_+a_.*Sqrt[w_]),x_Symbol] :=
  Int[u*v/(v^2-a^2*w),x] -
  Dist[a,Int[u*Sqrt[w]/(v^2-a^2*w),x]] /;
FreeQ[a,x] && PolynomialQ[v,x] *)
```

```
(* Int[u_./(a_.*x_+b_.*Sqrt[c_+d_.*x_]),x_Symbol] :=
  Int[(a*x*u-b*u*Sqrt[c+d*x])/(-b^2*c-b^2*d*x+a^2*x^2),x] /;
FreeQ[{a,b,c,d},x] *)
```

$$\int \frac{u\sqrt{a^2 - b^2 x^2}}{a + b x}\, dx$$

- **Derivation: Algebraic simplification**
- **Basis:** $\frac{\sqrt{z^2-w^2}}{z+w} = \frac{z}{\sqrt{z^2-w^2}} - \frac{w}{\sqrt{z^2-w^2}}$
- **Rule: If** $m n + 1 = 0$**, then**

$$\int \frac{u\sqrt{a^2 - b^2 x^2}}{a + b x}\, dx \longrightarrow a \int \frac{u}{\sqrt{a^2 - b^2 x^2}}\, dx - b \int \frac{x u}{\sqrt{a^2 - b^2 x^2}}\, dx$$

- **Program code:**

```
Int[u_.*Sqrt[c_+d_.*x_^2]/(a_+b_.*x_),x_Symbol] :=
  a*Int[u/Sqrt[c+d*x^2],x] -
  b*Int[x*u/Sqrt[c+d*x^2],x] /;
FreeQ[{a,b,c,d},x] && ZeroQ[c-a^2] && ZeroQ[d+b^2]
```

$$\int \frac{\sqrt{b x^2 + \sqrt{a + b^2 x^4}}}{\sqrt{a + b^2 x^4}} \, dx$$

- **Rule: If $b > 0$, then**

$$\int \frac{\sqrt{b x^2 + \sqrt{a + b^2 x^4}}}{\sqrt{a + b^2 x^4}} \, dx \longrightarrow \frac{1}{\sqrt{2 b}} \operatorname{ArcTanh}\left[\frac{\sqrt{2 b} \, x}{\sqrt{b x^2 + \sqrt{a + b^2 x^4}}}\right]$$

- **Program code:**

```
Int[Sqrt[b_.*x_^2+Sqrt[a_+c_.*x_^4]]/Sqrt[a_+c_.*x_^4],x_Symbol] :=
  ArcTanh[Rt[2*b,2]*x/Sqrt[b*x^2+Sqrt[a+c*x^4]]]/Rt[2*b,2] /;
FreeQ[{a,b,c},x] && ZeroQ[c-b^2] && PosQ[b]
```

- **Rule: If $\neg (b > 0)$, then**

$$\int \frac{\sqrt{b x^2 + \sqrt{a + b^2 x^4}}}{\sqrt{a + b^2 x^4}} \, dx \longrightarrow \frac{1}{\sqrt{-2 b}} \operatorname{ArcTan}\left[\frac{\sqrt{-2 b} \, x}{\sqrt{b x^2 + \sqrt{a + b^2 x^4}}}\right]$$

- **Program code:**

```
Int[Sqrt[b_.*x_^2+Sqrt[a_+c_.*x_^4]]/Sqrt[a_+c_.*x_^4],x_Symbol] :=
  ArcTan[Rt[-2*b,2]*x/Sqrt[b*x^2+Sqrt[a+c*x^4]]]/Rt[-2*b,2] /;
FreeQ[{a,b,c},x] && ZeroQ[c-b^2] && NegQ[b]
```

$$\int \frac{u \sqrt{v + \sqrt{a + v^2}}}{\sqrt{a + v^2}} \, dx$$

- **Author: Martin**
- **Derivation: Algebraic expansion**
- **Basis: If** $a > 0$**, then** $\sqrt{a + z^2} = \sqrt{\sqrt{a} + i z} \sqrt{\sqrt{a} - i z}$
- **Basis: If** $a > 0$**, then** $\frac{\sqrt{z + \sqrt{a + z^2}}}{\sqrt{a + z^2}} = \frac{1 - i}{2 \sqrt{\sqrt{a} - i z}} + \frac{1 + i}{2 \sqrt{\sqrt{a} + i z}}$
- **Rule: If** $a > 0$**, then**

$$\int \frac{u \sqrt{v + \sqrt{a + v^2}}}{\sqrt{a + v^2}} \, dx \longrightarrow \frac{1 - i}{2} \int \frac{u}{\sqrt{\sqrt{a} - i v}} \, dx + \frac{1 + i}{2} \int \frac{u}{\sqrt{\sqrt{a} + i v}} \, dx$$

- **Program code:**

```
Int[u_.*Sqrt[v_+Sqrt[a_+w_]]/Sqrt[a_+w_],x_Symbol] :=
  Dist[(1-I)/2, Int[u/Sqrt[Sqrt[a]-I*v],x]] +
  Dist[(1+I)/2, Int[u/Sqrt[Sqrt[a]+I*v],x]] /;
FreeQ[a,x] && ZeroQ[w-v^2] && PositiveQ[a] && Not[LinearQ[v,x]]
```

$$\int \frac{1}{a\,c + b\,c\,u}\,dx$$

- **Note: Constant factors in denominator are aggressively factored out to prevent them occurring unnecessarily in logarithm of antiderivative!**
- **Rule:**

$$\int \frac{1}{a\,c + b\,c\,u}\,dx \;\longrightarrow\; \frac{1}{c}\int \frac{1}{a + b\,u}\,dx$$

- **Program code:**

```
If[ShowSteps,

Int[1/(a_+b_.*u_),x_Symbol] :=
  Module[{lst=ConstantFactor[a+b*u,x]},
  ShowStep["","Int[1/(a*c+b*c*u),x]","Int[1/(a+b*u),x]/c",Hold[
  Dist[1/lst[[1]],Int[1/lst[[2]],x]]]]] /;
 lst[[1]]=!=1] /;
SimplifyFlag && FreeQ[{a,b},x] && (
    MatchQ[u,f_^(c_.+d_.*x) /; FreeQ[{c,d,f},x]] ||
    MatchQ[u,f_[c_.+d_.*x] /; FreeQ[{c,d},x] && MemberQ[{Tan,Cot,Tanh,Coth},f]]),

Int[1/(a_+b_.*u_),x_Symbol] :=
  Module[{lst=ConstantFactor[a+b*u,x]},
  Dist[1/lst[[1]],Int[1/lst[[2]],x]] /;
 lst[[1]]=!=1] /;
FreeQ[{a,b},x] && (
    MatchQ[u,f_^(c_.+d_.*x) /; FreeQ[{c,d,f},x]] ||
    MatchQ[u,f_[c_.+d_.*x] /; FreeQ[{c,d},x] && MemberQ[{Tan,Cot,Tanh,Coth},f]])]
```