$$\int \frac{1}{a + b\,x + c\,x^2}\,dx$$

- **Reference: G&R 2.172.2, CRC 110a, A&S 3.3.17**
- **Derivation: Integration by substitution**
- **Basis:** $\frac{1}{a+b\,x+c\,x^2} = -\frac{2}{\sqrt{b^2-4\,a\,c}}\,\frac{1}{1-\left(\frac{b+2\,c\,x}{\sqrt{b^2-4\,a\,c}}\right)^2}\,\partial_x\left(\frac{b+2\,c\,x}{\sqrt{b^2-4\,a\,c}}\right)$
- **Rule: If** $b^2-4\,a\,c>0$, **then**

$$\int \frac{1}{a+b\,x+c\,x^2}\,dx \longrightarrow -\frac{2}{\sqrt{b^2-4\,a\,c}}\,\text{ArcTanh}\left[\frac{b+2\,c\,x}{\sqrt{b^2-4\,a\,c}}\right]$$

- **Program code:**

```
Int[1/(a_+b_.*x_+c_.*x_^2),x_Symbol] :=
  Module[{q=Rt[b^2-4*a*c,2]},
  -2*ArcTanh[b/q+2*c*x/q]/q /;
 RationalQ[q] || SqrtNumberQ[q] && RationalQ[b/q]] /;
FreeQ[{a,b,c},x] && PosQ[b^2-4*a*c]
```

```
Int[1/(a_+b_.*x_+c_.*x_^2),x_Symbol] :=
  -2*ArcTanh[(b+2*c*x)/Rt[b^2-4*a*c,2]]/Rt[b^2-4*a*c,2] /;
FreeQ[{a,b,c},x] && PosQ[b^2-4*a*c]
```

- **Reference: G&R 2.172.4, CRC 109, A&S 3.3.16**
- **Derivation: Integration by substitution**
- **Basis:** $\frac{1}{a+b\,x+c\,x^2} = \frac{2}{\sqrt{4\,a\,c-b^2}}\,\frac{1}{1+\left(\frac{b+2\,c\,x}{\sqrt{4\,a\,c-b^2}}\right)^2}\,\partial_x\left(\frac{b+2\,c\,x}{\sqrt{4\,a\,c-b^2}}\right)$
- **Rule: If** $\neg\,(b^2-4\,a\,c>0)$, **then**

$$\int \frac{1}{a+b\,x+c\,x^2}\,dx \longrightarrow \frac{2}{\sqrt{4\,a\,c-b^2}}\,\text{ArcTan}\left[\frac{b+2\,c\,x}{\sqrt{4\,a\,c-b^2}}\right]$$

- **Program code:**

```
Int[1/(a_+b_.*x_+c_.*x_^2),x_Symbol] :=
  Module[{q=Rt[4*a*c-b^2,2]},
  2*ArcTan[b/q+2*c*x/q]/q /;
 RationalQ[q] || SqrtNumberQ[q] && RationalQ[b/q]] /;
FreeQ[{a,b,c},x] && NegQ[b^2-4*a*c]
```

```
Int[1/(a_+b_.*x_+c_.*x_^2),x_Symbol] :=
  2*ArcTan[(b+2*c*x)/Rt[4*a*c-b^2,2]]/Rt[4*a*c-b^2,2] /;
FreeQ[{a,b,c},x] && NegQ[b^2-4*a*c]
```

$$\int \left(a+b\,x+c\,x^2\right)^n \,dx$$

- **Derivation: Algebraic simplification**
- **Basis: If** $b^2-4\,a\,c=0$, **then** $a+b\,z+c\,z^2=\frac{1}{c}\left(\frac{b}{2}+c\,z\right)^2$
- **Rule: If** $n\in\mathbb{Z}\wedge b^2-4\,a\,c=0$, **then**

$$\int \left(a+b\,x+c\,x^2\right)^n \,dx \;\longrightarrow\; \frac{1}{c^n}\int\left(\frac{b}{2}+c\,x\right)^{2\,n}dx$$

- **Program code:**

```
Int[(a_+b_.*x_+c_.*x_^2)^n_,x_Symbol] :=
  Int[(b/2+c*x)^(2*n),x]/c^n /;
FreeQ[{a,b,c},x] && IntegerQ[n] && ZeroQ[b^2-4*a*c]
```

- **Reference: G&R 2.171.3, GR5 2.263.3, CRC 113**
- **Rule: If** $n\in\mathbb{Z}\wedge n<-1\wedge b^2-4\,a\,c\neq 0$, **then**

$$\int \left(a+b\,x+c\,x^2\right)^n \,dx \;\longrightarrow\; \frac{(b+2\,c\,x)\left(a+b\,x+c\,x^2\right)^{n+1}}{(n+1)\left(b^2-4\,a\,c\right)}-\frac{2\,c\,(2\,n+3)}{(n+1)\left(b^2-4\,a\,c\right)}\int\left(a+b\,x+c\,x^2\right)^{n+1}dx$$

- **Program code:**

```
Int[(a_+b_.*x_+c_.*x_^2)^n_,x_Symbol] :=
  (b+2*c*x)*(a+b*x+c*x^2)^(n+1)/((n+1)*(b^2-4*a*c)) -
  Dist[2*c*(2*n+3)/((n+1)*(b^2-4*a*c)),Int[(a+b*x+c*x^2)^(n+1),x]] /;
FreeQ[{a,b,c},x] && IntegerQ[n] && n<-1 && NonzeroQ[b^2-4*a*c]
```

$$\int \frac{d + e x}{a + b x + c x^2} \, dx$$

- **Reference: G&R 2.175.1, CRC 114**
- **Rule: If $2 c d - b e = 0$, then**

$$\int \frac{d + e x}{-a + b x + c x^2} \, dx \longrightarrow \frac{e \operatorname{Log}\left[a - b x - c x^2\right]}{2 c}$$

- **Program code:**

```
Int[(d_.+e_.*x_)/(a_+b_.*x_+c_.*x_^2),x_Symbol] :=
  e*Log[-a-b*x-c*x^2]/(2*c) /;
FreeQ[{a,b,c,d,e},x] && ZeroQ[2*c*d-b*e] && NegativeCoefficientQ[a]
```

- **Reference: G&R 2.175.1, CRC 114**
- **Rule: If $2 c d - b e = 0$, then**

$$\int \frac{d + e x}{a + b x + c x^2} \, dx \longrightarrow \frac{e \operatorname{Log}\left[a + b x + c x^2\right]}{2 c}$$

- **Program code:**

```
Int[(d_.+e_.*x_)/(a_+b_.*x_+c_.*x_^2),x_Symbol] :=
  e*Log[a+b*x+c*x^2]/(2*c) /;
FreeQ[{a,b,c,d,e},x] && ZeroQ[2*c*d-b*e]
```

- **Reference: A&S 3.3.19**
- **Rule: If $\sqrt{b^2 - 4 a c} \notin \mathbb{Q} \bigwedge a e^2 + c d^2 - b d e \neq 0$, then**

$$\int \frac{d + e x}{-a + b x + c x^2} \, dx \longrightarrow \frac{e \operatorname{Log}\left[a - b x - c x^2\right]}{2 c} + \frac{2 c d - b e}{2 c} \int \frac{1}{-a + b x + c x^2} \, dx$$

- **Program code:**

```
Int[(d_.+e_.*x_)/(a_+b_.*x_+c_.*x_^2),x_Symbol] :=
  e*Log[-a-b*x-c*x^2]/(2*c) +
  Dist[Simplify[(2*c*d-b*e)/(2*c)],Int[1/(a+b*x+c*x^2),x]] /;
FreeQ[{a,b,c,d,e},x] && Not[RationalQ[Rt[b^2-4*a*c,2]]] && NonzeroQ[a*e^2+c*d^2-b*d*e] &&
NegativeCoefficientQ[a]
```

- **Reference: A&S 3.3.19**
- **Rule: If** $\sqrt{b^2 - 4\,a\,c} \notin \mathbb{Q} \bigwedge a\,e^2 + c\,d^2 - b\,d\,e \neq 0$**, then**

$$\int \frac{d + e\,x}{a + b\,x + c\,x^2}\,dx \longrightarrow \frac{e\,\mathrm{Log}\left[a + b\,x + c\,x^2\right]}{2\,c} + \frac{2\,c\,d - b\,e}{2\,c}\,\mathrm{Int}\left[\frac{1}{a + b\,x + c\,x^2},\,x\right]$$

- **Program code:**

```
Int[(d_.+e_.*x_)/(a_+b_.*x_+c_.*x_^2),x_Symbol] :=
  e*Log[a+b*x+c*x^2]/(2*c) +
  Dist[Simplify[(2*c*d-b*e)/(2*c)],Int[1/(a+b*x+c*x^2),x]] /;
FreeQ[{a,b,c,d,e},x] && Not[RationalQ[Rt[b^2-4*a*c,2]]] && NonzeroQ[a*e^2+c*d^2-b*d*e]
```

$$\int (d + e x)^m \left(a + b x + c x^2\right)^n \, dx$$

- **Derivation: Algebraic simplification**

- **Basis: If** $a e^2 - b d e + c d^2 = 0$**, then** $a + b x + c x^2 = (d + e x) \left(\frac{a}{d} + \frac{c x}{e}\right)$

- **Rule: If** $a e^2 - b d e + c d^2 = 0 \land n \in \mathbb{Z}$**, then**

$$\int (d + e x)^m \left(a + b x + c x^2\right)^n dx \longrightarrow \int (d + e x)^{m+n} \left(\frac{a}{d} + \frac{c x}{e}\right)^n dx$$

- **Program code:**

```
Int[(d_+e_.*x_)^m_.*(a_.+b_.*x_+c_.*x_^2)^n_.,x_Symbol] :=
  Int[(d+e*x)^(m+n)*(a/d+c/e*x)^n,x] /;
FreeQ[{a,b,c,d,e,m},x] && ZeroQ[a*e^2-b*d*e+c*d^2] && IntegerQ[n]
```

```
Int[(d_+e_.*x_)^m_.*(a_+c_.*x_^2)^n_.,x_Symbol] :=
  Int[(d+e*x)^(m+n)*(a/d+c/e*x)^n,x] /;
FreeQ[{a,c,d,e,m},x] && ZeroQ[a*e^2+c*d^2] && IntegerQ[n]
```

- **Rule: If** $a e^2 + c d^2 = 0 \land m + 2 (n + 1) = 0 \land n \notin \mathbb{Z}$**, then**

$$\int (d + e x)^m \left(a + c x^2\right)^n dx \longrightarrow \frac{e (d + e x)^m \left(a + c x^2\right)^{n+1}}{2 c d (n + 1)}$$

- **Program code:**

```
Int[(d_+e_.*x_)^m_*(a_+c_.*x_^2)^n_,x_Symbol] :=
  e*(d+e*x)^m*(a+c*x^2)^(n+1)/(2*c*d*(n+1)) /;
FreeQ[{a,c,d,e,m,n},x] && ZeroQ[a*e^2+c*d^2] && ZeroQ[m+2*(n+1)] && Not[IntegerQ[n]]
```

- **Rule: If** $a e^2 + c d^2 = 0 \land m + n + 1 \neq 0 \land m + 2 (n + 1) \neq 0 \land m < -1 \land n \notin \mathbb{Z}$**, then**

$$\int (d + e x)^m \left(a + c x^2\right)^n dx \longrightarrow -\frac{e (d + e x)^m \left(a + c x^2\right)^{n+1}}{2 c d (m + n + 1)} + \frac{m + 2 (n + 1)}{2 d (m + n + 1)} \int (d + e x)^{m+1} \left(a + c x^2\right)^n dx$$

- **Program code:**

```
Int[(d_+e_.*x_)^m_*(a_+c_.*x_^2)^n_,x_Symbol] :=
  -e*(d+e*x)^m*(a+c*x^2)^(n+1)/(2*c*d*(m+n+1)) +
  Dist[(m+2*(n+1))/(2*d*(m+n+1)),Int[(d+e*x)^(m+1)*(a+c*x^2)^n,x]] /;
FreeQ[{a,c,d,e,n},x] && ZeroQ[a*e^2+c*d^2] && NonzeroQ[m+n+1] && NonzeroQ[m+2*(n+1)] &&
  RationalQ[m] && m<-1 && Not[IntegerQ[n]]
```

- **Reference: G&R 2.174.2**

- **Rule: If** $n < -1 \land m + 2n + 1 = 0 \land 2cd - be = 0$**, then**

$$\int (d + e x)^m \left(a + b x + c x^2\right)^n dx \longrightarrow -\frac{e (d + e x)^{m-1} \left(a + b x + c x^2\right)^{n+1}}{c (m - 1)} + \frac{e^2}{c} \int (d + e x)^{m-2} \left(a + b x + c x^2\right)^{n+1} dx$$

- **Program code:**

```
Int[(d_.+e_.*x_)^m_*(a_.+b_.*x_+c_.*x_^2)^n_,x_Symbol] :=
  -e*(d+e*x)^(m-1)*(a+b*x+c*x^2)^(n+1)/(c*(m-1)) +
  Dist[e^2/c,Int[(d+e*x)^(m-2)*(a+b*x+c*x^2)^(n+1),x]] /;
FreeQ[{a,b,c,d,e},x] && RationalQ[{m,n}] && n<-1 && ZeroQ[m+2*n+1] && ZeroQ[2*c*d-b*e]
```

- **Reference: G&R 2.174.1, CRC 119**

- **Rule: If** $2cd - be = 0 \land n + 1 \neq 0$**, then**

$$\int (d + e x) \left(a + b x + c x^2\right)^n dx \longrightarrow \frac{e \left(a + b x + c x^2\right)^{n+1}}{2 c (n + 1)}$$

- **Program code:**

```
Int[(d_.+e_.*x_)*(a_.+b_.*x_+c_.*x_^2)^n_,x_Symbol] :=
  e*(a+b*x+c*x^2)^(n+1)/(2*c*(n+1)) /;
FreeQ[{a,b,c,d,e,n},x] && ZeroQ[2*c*d-b*e] && NonzeroQ[n+1]
```

- **Reference: G&R 2.174.1, CRC 119**

- **Rule: If** $n + 1 \neq 0 \land \neg (n \in \mathbb{Z} \land n > 0) \land 2cd - be \neq 0$**, then**

$$\int (d + e x) \left(a + b x + c x^2\right)^n dx \longrightarrow \frac{e \left(a + b x + c x^2\right)^{n+1}}{2 c (n + 1)} + \frac{2 c d - b e}{2 c} \int \left(a + b x + c x^2\right)^n dx$$

- **Program code:**

```
Int[(d_.+e_.*x_)*(a_.+b_.*x_+c_.*x_^2)^n_,x_Symbol] :=
  e*(a+b*x+c*x^2)^(n+1)/(2*c*(n+1)) +
  Dist[(2*c*d-b*e)/(2*c),Int[(a+b*x+c*x^2)^n,x]] /;
FreeQ[{a,b,c,d,e,n},x] && NonzeroQ[n+1] && Not[IntegerQ[n] && n>0] && NonzeroQ[2*c*d-b*e]
```

```
Int[(d_+e_.*x_)*(a_.+c_.*x_^2)^n_,x_Symbol] :=
  e*(a+c*x^2)^(n+1)/(2*c*(n+1)) +
  Dist[d,Int[(a+c*x^2)^n,x]] /;
FreeQ[{a,c,d,e,n},x] && NonzeroQ[n+1] && Not[IntegerQ[n] && n>0]
```

- **Reference: G&R 2.174.1, CRC 119**
- **Rule: If** $m>1 \land m+2n+1\neq 0 \land \neg (n\in\mathbb{Z} \land n\geq -1) \land (m+n=0 \lor 2cd-be=0)$, **then**

$$\int (d+e x)^m \left(a+b x+c x^2\right)^n dx \longrightarrow$$
$$\frac{e (d+e x)^{m-1} \left(a+b x+c x^2\right)^{n+1}}{c (m+2n+1)}-\frac{\left(a e^2-b d e+c d^2\right) (m-1)}{c (m+2n+1)}\int (d+e x)^{m-2} \left(a+b x+c x^2\right)^n dx$$

- **Program code:**

```
Int[(d_.+e_.*x_)^m_*(a_.+b_.*x_+c_.*x_^2)^n_,x_Symbol] :=
  e*(d+e*x)^(m-1)*(a+b*x+c*x^2)^(n+1)/(c*(m+2*n+1)) -
  Dist[(a*e^2-b*d*e+c*d^2)*(m-1)/(c*(m+2*n+1)),Int[(d+e*x)^(m-2)*(a+b*x+c*x^2)^n,x]] /;
FreeQ[{a,b,c,d,e,n},x] && RationalQ[m] && m>1 && NonzeroQ[m+2*n+1] && Not[IntegerQ[n] && n≥-1] &&
(ZeroQ[m+n] || ZeroQ[2*c*d-b*e])
```

```
Int[(d_+e_.*x_)^m_*(a_.+c_.*x_^2)^n_,x_Symbol] :=
  -e*(d+e*x)^(m-1)/(c*(m-1)*(a+c*x^2)^(m-1)) +
  Dist[(a*e^2+c*d^2)/c,Int[(d+e*x)^(m-2)/(a+c*x^2)^m,x]] /;
FreeQ[{a,c,d,e,n},x] && RationalQ[m] && m>1 && ZeroQ[m+n]
```

- **Reference: G&R 2.174.1, CRC 119**
- **Rule: If** $m>1 \land m+2n+1\neq 0 \land \neg (n\in\mathbb{Z} \land n\geq -1) \land a e^2-b d e+c d^2=0$, **then**

$$\int (d+e x)^m \left(a+b x+c x^2\right)^n dx \longrightarrow$$
$$\frac{e (d+e x)^{m-1} \left(a+b x+c x^2\right)^{n+1}}{c (m+2n+1)}+\frac{(2 c d-b e) (m+n)}{c (m+2n+1)}\int (d+e x)^{m-1} \left(a+b x+c x^2\right)^n dx$$

- **Program code:**

```
Int[(d_.+e_.*x_)^m_.*(a_.+b_.*x_+c_.*x_^2)^n_,x_Symbol] :=
  e*(d+e*x)^(m-1)*(a+b*x+c*x^2)^(n+1)/(c*(m+2*n+1)) +
  Dist[(2*c*d-b*e)*(m+n)/(c*(m+2*n+1)),Int[(d+e*x)^(m-1)*(a+b*x+c*x^2)^n,x]] /;
FreeQ[{a,b,c,d,e,n},x] && RationalQ[m] && m>1 && NonzeroQ[m+2*n+1] && Not[IntegerQ[n] && n>=-1] &&
ZeroQ[a*e^2-b*d*e+c*d^2]
```

```
Int[(d_+e_.*x_)^m_.*(a_.+c_.*x_^2)^n_,x_Symbol] :=
  e*(d+e*x)^(m-1)*(a+c*x^2)^(n+1)/(c*(m+2*n+1)) +
  Dist[2*c*d*(m+n)/(c*(m+2*n+1)),Int[(d+e*x)^(m-1)*(a+c*x^2)^n,x]] /;
FreeQ[{a,c,d,e,n},x] && RationalQ[m] && m>1 && NonzeroQ[m+2*n+1] && Not[IntegerQ[n] && n>=-1] &&
ZeroQ[a*e^2+c*d^2]
```

- **Reference:** G&R 2.265c special case

- **Rule:** If $m+n+1 \neq 0 \wedge m+2(n+1)=0$, then

$$\int x^m \left(b x+c x^2\right)^n dx \longrightarrow \frac{x^m \left(b x+c x^2\right)^{n+1}}{b (m+n+1)}$$

- **Program code:**

```
Int[x_^m_*(b_.*x_+c_.*x_^2)^n_,x_Symbol] :=
  x^m*(b*x+c*x^2)^(n+1)/(b*(m+n+1)) /;
FreeQ[{b,c,m,n},x] && NonzeroQ[m+n+1] && ZeroQ[m+2*(n+1)]
```

- **Reference:** G&R 2.265c

- **Rule:** If $m<-1 \wedge m+n+1 \neq 0 \wedge \neg (n \in \mathbb{Z} \wedge n \geq -1)$, then

$$\int x^m \left(b x+c x^2\right)^n dx \longrightarrow \frac{x^m \left(b x+c x^2\right)^{n+1}}{b (m+n+1)}-\frac{c (m+2 (n+1))}{b (m+n+1)} \int x^{m+1} \left(b x+c x^2\right)^n dx$$

- **Program code:**

```
Int[x_^m_*(b_.*x_+c_.*x_^2)^n_,x_Symbol] :=
  x^m*(b*x+c*x^2)^(n+1)/(b*(m+n+1)) -
  Dist[c*(m+2*(n+1))/(b*(m+n+1)),Int[x^(m+1)*(b*x+c*x^2)^n,x]] /;
FreeQ[{b,c,n},x] && RationalQ[m] && m<-1 && NonzeroQ[m+n+1] && Not[IntegerQ[n] && n≥-1]
```

- **Reference:** G&R 2.176, CRC 123

- **Rule:** If $m<-1 \wedge a e^2-b d e+c d^2 \neq 0 \wedge \neg (n \in \mathbb{Z} \wedge n \geq -1) \wedge m+2 n+3=0$, then

$$\int (d+e x)^m \left(a+b x+c x^2\right)^n dx \longrightarrow$$

$$\frac{e (d+e x)^{m+1} \left(a+b x+c x^2\right)^{n+1}}{(m+1) \left(a e^2-b d e+c d^2\right)}+\frac{(2 c d-b e) (m+n+2)}{(m+1) \left(a e^2-b d e+c d^2\right)} \int (d+e x)^{m+1} \left(a+b x+c x^2\right)^n dx$$

- **Program code:**

```
Int[(d_.+e_.*x_)^m_*(a_+b_.*x_+c_.*x_^2)^n_,x_Symbol] :=
  e*(d+e*x)^(m+1)*(a+b*x+c*x^2)^(n+1)/((m+1)*(a*e^2-b*d*e+c*d^2)) +
  Dist[(2*c*d-b*e)*(m+n+2)/((m+1)*(a*e^2-b*d*e+c*d^2)),Int[(d+e*x)^(m+1)*(a+b*x+c*x^2)^n,x]] /;
FreeQ[{a,b,c,d,e,n},x] && RationalQ[m] && m<-1 && NonzeroQ[a*e^2-b*d*e+c*d^2] &&
Not[IntegerQ[n] && n≥-1] && ZeroQ[m+2*n+3]
```

```
Int[(d_+e_.*x_)^m_*(a_+c_.*x_^2)^n_,x_Symbol] :=
  e*(d+e*x)^(m+1)*(a+c*x^2)^(n+1)/((m+1)*(a*e^2+c*d^2)) +
  Dist[2*c*d*(m+n+2)/((m+1)*(a*e^2+c*d^2)),Int[(d+e*x)^(m+1)*(a+c*x^2)^n,x]] /;
FreeQ[{a,c,d,e,n},x] && RationalQ[m] && m<-1 && NonzeroQ[a*e^2+c*d^2] &&
Not[IntegerQ[n] && n≥-1] && ZeroQ[m+2*n+3]
```

$$\int \frac{1}{a+b\,x^2+c\,x^4}\,dx$$

- **Derivation: Algebraic expansion**
- **Basis: Let** $q=\sqrt{\frac{a}{c}}$**, then** $\frac{1}{a+b\,x^2+c\,x^4}=\frac{c\,q\,(q+x^2)}{2\,a\,(a+b\,x^2+c\,x^4)}+\frac{c\,q\,(q-x^2)}{2\,a\,(a+b\,x^2+c\,x^4)}$
- **Note: Although resulting integrands appear more complicated than the original one, they are of the form required for the first two rules in the next section.**
- **Rule: If** $b^2-4\,a\,c\neq0\,\bigwedge\,\frac{a}{c}>0\,\bigwedge\,\left(b^2-4\,a\,c<0\,\bigvee\,\left(\frac{a}{c}\in\mathbb{Q}\,\bigwedge\,\neg\,(b^2-4\,a\,c>0)\right)\right)$**, let** $q=\sqrt{\frac{a}{c}}$**, then**

$$\int\frac{1}{a+b\,x^2+c\,x^4}\,dx\;\longrightarrow\;\frac{c\,q}{2\,a}\int\frac{q+x^2}{a+b\,x^2+c\,x^4}\,dx+\frac{c\,q}{2\,a}\int\frac{q-x^2}{a+b\,x^2+c\,x^4}\,dx$$

- **Program code:**

```
Int[1/(a_+b_.*x_^2+c_.*x_^4), x_Symbol] :=
  Module[{q=Rt[a/c,2]},
  Dist[c*q/(2*a),Int[(q+x^2)/(a+b*x^2+c*x^4),x]] +
  Dist[c*q/(2*a),Int[(q-x^2)/(a+b*x^2+c*x^4),x]]] /;
FreeQ[{a,b,c},x] && NonzeroQ[b^2-4*a*c] && PosQ[a/c] &&
(NegativeQ[b^2-4*a*c] || RationalQ[a/c] && Not[PositiveQ[b^2-4*a*c]])
```

- **Derivation: Algebraic expansion**
- **Basis: Let** $q=\sqrt{-\frac{a}{c}}$**, then** $\frac{1}{a+b\,x^2+c\,x^4}=-\frac{c\,q\,(q+x^2)}{2\,a\,(a+b\,x^2+c\,x^4)}-\frac{c\,q\,(q-x^2)}{2\,a\,(a+b\,x^2+c\,x^4)}$
- **Note: Although resulting integrands appear more complicated than the original one, they are of the form required for the first two rules in the next section.**
- **Rule: If** $b^2-4\,a\,c\neq0\,\bigwedge\,\neg\left(\frac{a}{c}>0\right)\,\bigwedge\,\left(b^2-4\,a\,c<0\,\bigvee\,\left(\frac{a}{c}\in\mathbb{Q}\,\bigwedge\,\neg\,(b^2-4\,a\,c>0)\right)\right)$**, let** $q=\sqrt{-\frac{a}{c}}$**, then**

$$\int\frac{1}{a+b\,x^2+c\,x^4}\,dx\;\longrightarrow\;-\frac{c\,q}{2\,a}\int\frac{q+x^2}{a+b\,x^2+c\,x^4}\,dx-\frac{c\,q}{2\,a}\int\frac{q-x^2}{a+b\,x^2+c\,x^4}\,dx$$

- **Program code:**

```
Int[1/(a_+b_.*x_^2+c_.*x_^4), x_Symbol] :=
  Module[{q=Rt[-a/c,2]},
  -Dist[c*q/(2*a),Int[(q+x^2)/(a+b*x^2+c*x^4),x]] -
  Dist[c*q/(2*a),Int[(q-x^2)/(a+b*x^2+c*x^4),x]]] /;
FreeQ[{a,b,c},x] && NonzeroQ[b^2-4*a*c] && NegQ[a/c] &&
(NegativeQ[b^2-4*a*c] || RationalQ[a/c] && Not[PositiveQ[b^2-4*a*c]])
```

$$\int \frac{d+e\,x^2}{a+b\,x^2+c\,x^4}\,dx$$

- **Note: Previously undiscovered rule?**

- **Rule: If** $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-a\,e^2=0\,\bigwedge\,\neg\left(\frac{b\,d+2\,a\,e}{a\,d}>0\right)$**, then**

$$\int \frac{d+e\,x^2}{a+b\,x^2+c\,x^4}\,dx\,\longrightarrow\,\frac{d}{a\sqrt{-\frac{b\,d+2\,a\,e}{a\,d}}}\,\text{ArcTanh}\left[\frac{d\sqrt{-\frac{b\,d+2\,a\,e}{a\,d}}\,x}{d-e\,x^2}\right]$$

- **Program code:**

```
Int[(d_+e_.*x_^2)/(a_+b_.*x_^2+c_.*x_^4), x_Symbol] :=
  d/(a*Rt[-(b*d+2*a*e)/(a*d),2])*ArcTanh[d*Rt[-(b*d+2*a*e)/(a*d),2]*x/(d-e*x^2)] /;
FreeQ[{a,b,c,d,e},x] && NonzeroQ[b^2-4*a*c] && ZeroQ[c*d^2-a*e^2] && NegQ[(b*d+2*a*e)/(a*d)]
```

- **Note: Although this rule would produce superficially simpler antiderivatives than the following rule, unfortunately they are discontinuous at the points** $x=\frac{a}{c}$ **and** $x=-\frac{a}{c}$**.**

- **Rule: If** $b^2-4\,a\,c\neq 0\,\wedge\,c\,d^2-a\,e^2=0\,\bigwedge\,\frac{b\,d+2\,a\,e}{a\,d}>0$**, then**

$$\int \frac{d+e\,x^2}{a+b\,x^2+c\,x^4}\,dx\,\longrightarrow\,\frac{d}{a\sqrt{\frac{b\,d+2\,a\,e}{a\,d}}}\,\text{ArcTan}\left[\frac{d\sqrt{\frac{b\,d+2\,a\,e}{a\,d}}\,x}{d-e\,x^2}\right]$$

- **Program code:**

```
(* Int[(d_+e_.*x_^2)/(a_+b_.*x_^2+c_.*x_^4), x_Symbol] :=
  d/(a*Rt[(b*d+2*a*e)/(a*d),2])*ArcTan[d*Rt[(b*d+2*a*e)/(a*d),2]*x/(d-e*x^2)] /;
FreeQ[{a,b,c,d,e},x] && NonzeroQ[b^2-4*a*c] && ZeroQ[c*d^2-a*e^2] && PosQ[(b*d+2*a*e)/(a*d)] *)
```

- **Derivation: Algebraic expansion**

- **Basis: If** $c\,d^2 - a\,e^2 = 0$**, let** $q = \frac{c\,d}{e}$ **and** $r = \sqrt{2\,c\,q - b\,c}$**, then** $\frac{d+e\,x^2}{a+b\,x^2+c\,x^4} = \frac{e}{2\,(q+r\,x+c\,x^2)} + \frac{e}{2\,(q-r\,x+c\,x^2)}$

- **Rule: If** $b^2 - 4\,a\,c \neq 0 \wedge c\,d^2 - a\,e^2 = 0$**, let** $q = \frac{c\,d}{e}$**, if** $2\,c\,q - b\,c > 0$**, let** $r = \sqrt{2\,c\,q - b\,c}$**, then**

$$\int \frac{d+e\,x^2}{a+b\,x^2+c\,x^4}\,dx \to \frac{e}{2}\int \frac{1}{q-r\,x+c\,x^2}\,dx + \frac{e}{2}\int \frac{1}{q+r\,x+c\,x^2}\,dx$$

- **Program code:**

```
Int[(d_+e_.*x_^2)/(a_+b_.*x_^2+c_.*x_^4), x_Symbol] :=
  Module[{q=c*d/e},
  Module[{r=Rt[2*c*q-b*c,2]},
  Dist[e/2,Int[1/(q-r*x+c*x^2),x]] +
  Dist[e/2,Int[1/(q+r*x+c*x^2),x]]] /;
 Not[NegativeQ[2*c*q-b*c]]] /;
FreeQ[{a,b,c,d,e},x] && NonzeroQ[b^2-4*a*c] && ZeroQ[c*d^2-a*e^2] && PosQ[(b*d+2*a*e)/(a*d)]
```

- **Derivation: Algebraic expansion**

- **Basis: Let** $q = \sqrt{\frac{a}{c}}$**, then** $\frac{d+e\,x^2}{a+b\,x^2+c\,x^4} = \frac{(q\,c\,d+a\,e)\,(q+x^2)}{2\,a\,(a+b\,x^2+c\,x^4)} + \frac{(q\,c\,d-a\,e)\,(q-x^2)}{2\,a\,(a+b\,x^2+c\,x^4)}$

- **Note: Resulting integrands are of the form of the first two rules in this section.**

- **Rule: If** $b^2 - 4\,a\,c \neq 0 \bigwedge c\,d^2 - a\,e^2 \neq 0 \bigwedge \frac{a}{c} > 0 \bigwedge \left(b^2 - 4\,a\,c < 0 \bigvee \left(\frac{a}{c} \in \mathbb{Q} \bigwedge \neg\,(b^2 - 4\,a\,c > 0)\right)\right)$**, let** $q = \sqrt{\frac{a}{c}}$**, then**

$$\int \frac{d+e\,x^2}{a+b\,x^2+c\,x^4}\,dx \to \frac{q\,c\,d+a\,e}{2\,a}\int \frac{q+x^2}{a+b\,x^2+c\,x^4}\,dx + \frac{q\,c\,d-a\,e}{2\,a}\int \frac{q-x^2}{a+b\,x^2+c\,x^4}\,dx$$

- **Program code:**

```
Int[(d_.+e_.*x_^2)/(a_+b_.*x_^2+c_.*x_^4), x_Symbol] :=
  Module[{q=Rt[a/c,2]},
  Dist[(q*c*d+a*e)/(2*a),Int[(q+x^2)/(a+b*x^2+c*x^4),x]] +
  Dist[(q*c*d-a*e)/(2*a),Int[(q-x^2)/(a+b*x^2+c*x^4),x]]] /;
FreeQ[{a,b,c,d,e},x] && NonzeroQ[b^2-4*a*c] && NonzeroQ[c*d^2-a*e^2] && PosQ[a/c] &&
(NegativeQ[b^2-4*a*c] || RationalQ[a/c] && Not[PositiveQ[b^2-4*a*c]])
```

- **Derivation: Algebraic expansion**

- **Basis: Let** $q=\sqrt{-\frac{a}{c}}$ **, then** $\frac{d+e\,x^2}{a+b\,x^2+c\,x^4}=-\frac{(q\,c\,d-a\,e)\left(q+x^2\right)}{2\,a\left(a+b\,x^2+c\,x^4\right)}-\frac{(q\,c\,d+a\,e)\left(q-x^2\right)}{2\,a\left(a+b\,x^2+c\,x^4\right)}$

- **Note: Resulting integrands are of the form of the first two rules in this section.**

- **Rule: If** $b^2-4\,a\,c\neq 0\,\bigwedge\,c\,d^2-a\,e^2\neq 0\bigwedge\,\neg\left(\frac{a}{c}>0\right)\,\bigwedge\,\left(b^2-4\,a\,c<0\,\bigvee\,\left(\frac{a}{c}\in\mathbb{Q}\,\bigwedge\,\neg\,(b^2-4\,a\,c>0)\right)\right)$**, let** $q=\sqrt{-\frac{a}{c}}$ **,**

  then

$$\int\frac{d+e\,x^2}{a+b\,x^2+c\,x^4}\,dx\,\longrightarrow\,-\frac{q\,c\,d-a\,e}{2\,a}\int\frac{q+x^2}{a+b\,x^2+c\,x^4}\,dx-\frac{q\,c\,d+a\,e}{2\,a}\int\frac{q-x^2}{a+b\,x^2+c\,x^4}\,dx$$

- **Program code:**

```
Int[(d_+e_.*x_^2)/(a_+b_.*x_^2+c_.*x_^4), x_Symbol] :=
  Module[{q=Rt[-a/c,2]},
  Dist[-(q*c*d-a*e)/(2*a),Int[(q+x^2)/(a+b*x^2+c*x^4),x]] -
  Dist[(q*c*d+a*e)/(2*a),Int[(q-x^2)/(a+b*x^2+c*x^4),x]]] /;
FreeQ[{a,b,c,d,e},x] && NonzeroQ[b^2-4*a*c] && NonzeroQ[c*d^2-a*e^2] && NegQ[a/c] &&
(NegativeQ[b^2-4*a*c] || RationalQ[a/c] && Not[PositiveQ[b^2-4*a*c]])
```

$$\int \frac{1}{a + b\,x^n + c\,x^{2\,n}}\, dx$$

- **Reference: G&R 2.161.1b?**
- **Derivation: Algebraic expansion**
- **Basis: Let** $q = \sqrt{a\,c}$ **and** $r = \sqrt{2\,c\,q - b\,c}$, **then** $\frac{1}{a+b\,z^2+c\,z^4} = \frac{q}{2\,a\,r}\,\frac{r-c\,z}{q-r\,z+c\,z^2} + \frac{q}{2\,a\,r}\,\frac{r+c\,z}{q+r\,z+c\,z^2}$
- **Rule: If** $b^2 - 4\,a\,c < 0 \bigwedge \frac{n}{2} \in \mathbb{Z} \bigwedge n > 2 \bigwedge a\,c > 0$, **let** $q = \sqrt{a\,c}$, **if** $2\,c\,q - b\,c > 0$, **let** $r = \sqrt{2\,c\,q - b\,c}$, **then**

$$\int \frac{1}{a + b\,x^n + c\,x^{2\,n}}\, dx \;\longrightarrow\; \frac{q}{2\,a\,r}\int \frac{r - c\,x^{n/2}}{q - r\,x^{n/2} + c\,x^n}\, dx + \frac{q}{2\,a\,r}\int \frac{r + c\,x^{n/2}}{q + r\,x^{n/2} + c\,x^n}\, dx$$

- **Program code:**

```
Int[1/(a_+b_.*x_^n_+c_.*x_^j_),x_Symbol] :=
  Module[{q=Rt[a*c,2]},
  Module[{r=Rt[2*c*q-b*c,2]},
  Dist[q/(2*a*r),Int[(r-c*x^(n/2))/(q-r*x^(n/2)+c*x^n),x]] +
  Dist[q/(2*a*r),Int[(r+c*x^(n/2))/(q+r*x^(n/2)+c*x^n),x]]] /;
 Not[NegativeQ[2*c*q-b*c]]] /;
FreeQ[{a,b,c},x] && NegativeQ[b^2-4*a*c] && ZeroQ[j-2*n] && IntegerQ[n/2] && n>2 &&
  NegativeQ[b^2-4*a*c] && PosQ[a*c]
```

- **Reference: G&R 2.161.1a**
- **Derivation: Algebraic expansion**
- **Basis: Let** $q = \sqrt{b^2 - 4\,a\,c}$, **then** $\frac{1}{a+b\,z+c\,z^2} = \frac{2\,c}{q}\,\frac{1}{b-q+2\,c\,z} - \frac{2\,c}{q}\,\frac{1}{b+q+2\,c\,z}$
- **Rule: If** $b^2 - 4\,a\,c \neq 0 \bigwedge n \in \mathbb{Z} \bigwedge n > 1 \bigwedge \neg\left(\frac{n}{2} \in \mathbb{Z} \bigwedge b^2 - 4\,a\,c < 0 \bigwedge a\,c > 0\right)$, **let** $q = \sqrt{b^2 - 4\,a\,c}$, **then**

$$\int \frac{1}{a + b\,x^n + c\,x^{2\,n}}\, dx \;\longrightarrow\; \frac{2\,c}{q}\int \frac{1}{b - q + 2\,c\,x^n}\, dx - \frac{2\,c}{q}\int \frac{1}{b + q + 2\,c\,x^n}\, dx$$

- **Program code:**

```
Int[1/(a_+b_.*x_^n_+c_.*x_^j_),x_Symbol] :=
  Module[{q=Rt[b^2-4*a*c,2]},
  Dist[2*c/q,Int[1/(b-q+2*c*x^n),x]] -
  Dist[2*c/q,Int[1/(b+q+2*c*x^n),x]]] /;
FreeQ[{a,b,c},x] && NonzeroQ[b^2-4*a*c] && ZeroQ[j-2*n] && IntegerQ[n] && n>1 &&
  Not[IntegerQ[n/2] && NegativeQ[b^2-4*a*c] && PosQ[a*c]]
```

$$\int \frac{x^m}{a+b\,x^n+c\,x^{2\,n}}\,dx$$

- **Reference: G&R 2.177.1, CRC 120**
- **Derivation: Algebraic expansion**
- **Basis:** $\frac{1}{x\,(a+b\,x^n+c\,x^{2\,n})} = \frac{1}{a\,x} - \frac{1}{a}\,\frac{x^{n-1}\,(b+c\,x^n)}{a+b\,x^n+c\,x^{2\,n}}$
- **Note: Although the resulting integrand appears more complicated than the original one, it is easily integrated by the substitution $u = x^n$.**
- **Rule: If $b^2-4\,a\,c \neq 0 \wedge \neg\,(n<0)$, then**

$$\int \frac{1}{x\,(a+b\,x^n+c\,x^{2\,n})}\,dx \longrightarrow \frac{\mathrm{Log}[x]}{a} - \frac{1}{a}\int \frac{x^{n-1}\,(b+c\,x^n)}{a+b\,x^n+c\,x^{2\,n}}\,dx$$

- **Program code:**

```
Int[1/(x_*(a_+b_.*x_^n_+c_.*x_^j_)),x_Symbol] :=
  Log[x]/a - Dist[1/a,Int[x^(n-1)*(b+c*x^n)/(a+b*x^n+c*x^(2*n)),x]] /;
FreeQ[{a,b,c,n},x] && NonzeroQ[b^2-4*a*c] && ZeroQ[j-2*n] && Not[NegativeQ[n]]
```

- **Reference: G&R 2.176, CRC 123**
- **Derivation: Algebraic expansion**
- **Basis:** $\frac{x^m}{a+b\,x^n+c\,x^{2\,n}} = \frac{x^m}{a} - \frac{1}{a}\,\frac{x^{m+n}\,(b+c\,x^n)}{a+b\,x^n+c\,x^{2\,n}}$
- **Rule: If $b^2-4\,a\,c \neq 0 \wedge m,\,n \in \mathbb{Z} \wedge n>0 \wedge m<-1$, then**

$$\int \frac{x^m}{a+b\,x^n+c\,x^{2\,n}}\,dx \longrightarrow \frac{x^{m+1}}{a\,(m+1)} - \frac{1}{a}\int \frac{x^{m+n}\,(b+c\,x^n)}{a+b\,x^n+c\,x^{2\,n}}\,dx$$

- **Program code:**

```
Int[x_^m_/(a_+b_.*x_^n_+c_.*x_^j_),x_Symbol] :=
  x^(m+1)/(a*(m+1)) -
  Dist[1/a,Int[x^(m+n)*(b+c*x^n)/(a+b*x^n+c*x^(2*n)),x]] /;
FreeQ[{a,b,c},x] && NonzeroQ[b^2-4*a*c] && ZeroQ[j-2*n] && IntegersQ[m,n] && n>0 && m<-1
```

- **Reference: G&R 2.174.1, CRC 119**
- **Derivation: Algebraic expansion**
- **Basis:** $\frac{x^m}{a+b x^n+c x^{2n}} = \frac{x^{m-2n}}{c} - \frac{1}{c} \frac{x^{m-2n} (a+b x^n)}{a+b x^n+c x^{2n}}$
- **Rule: If** $b^2 - 4 a c \neq 0 \wedge m, 2n \in \mathbb{Z} \wedge 0 < 2n \leq m$, **then**

$$\int \frac{x^m}{a+b x^n+c x^{2n}} \, dx \longrightarrow \frac{x^{m-2n+1}}{c (m-2n+1)} - \frac{1}{c} \int \frac{x^{m-2n} (a+b x^n)}{a+b x^n+c x^{2n}} \, dx$$

- **Program code:**

```
Int[x_^m_./(a_+b_.*x_^n_+c_.*x_^j_),x_Symbol] :=
  x^(m-2*n+1)/(c*(m-2*n+1)) -
  Dist[1/c,Int[x^(m-2*n)*(a+b*x^n)/(a+b*x^n+c*x^(2*n)),x]] /;
FreeQ[{a,b,c},x] && NonzeroQ[b^2-4*a*c] && ZeroQ[j-2*n] && IntegersQ[m,n] && 0<2*n≤m
```

- **Derivation: Algebraic expansion**
- **Basis: Let** $q = \sqrt{a c}$ **and** $r = \sqrt{2 c q - b c}$, **then** $\frac{z^m}{a+b z^2+c z^4} = \frac{c}{2 r} \frac{z^{m-1}}{q-r z+c z^2} - \frac{c}{2 r} \frac{z^{m-1}}{q+r z+c z^2}$
- **Rule: If** $b^2 - 4 a c < 0 \wedge m, \frac{n}{2} \in \mathbb{Z} \bigwedge 1 < \frac{n}{2} \leq m < 2n \bigwedge \text{CoprimeQ}[m+1, n] \bigwedge a c > 0$, **let** $q = \sqrt{a c}$, **if** $2 c q - b c > 0$, **let** $r = \sqrt{2 c q - b c}$, **then**

$$\int \frac{x^m}{a+b x^n+c x^{2n}} \, dx \longrightarrow \frac{c}{2 r} \int \frac{x^{m-n/2}}{q-r x^{n/2}+c x^n} \, dx - \frac{c}{2 r} \int \frac{x^{m-n/2}}{q+r x^{n/2}+c x^n} \, dx$$

- **Program code:**

```
Int[x_^m_./(a_+b_.*x_^n_+c_.*x_^j_),x_Symbol] :=
  Module[{q=Rt[a*c,2]},
  Module[{r=Rt[2*c*q-b*c,2]},
  Dist[c/(2*r),Int[x^(m-n/2)/(q-r*x^(n/2)+c*x^n),x]] -
  Dist[c/(2*r),Int[x^(m-n/2)/(q+r*x^(n/2)+c*x^n),x]]] /;
 Not[NegativeQ[2*c*q-b*c]]] /;
FreeQ[{a,b,c},x] && NegativeQ[b^2-4*a*c] && ZeroQ[j-2*n] && IntegersQ[m,n/2] &&
  1<n/2≤m<2*n && CoprimeQ[m+1,n] && PosQ[a*c]
```

- **Reference: G&R 2.161.1a**
- **Derivation: Algebraic expansion**
- **Basis: Let** $q = \sqrt{b^2 - 4 a c}$**, then** $\frac{z^m}{a+b z+c z^2} = \frac{2 c}{q} \frac{z^m}{b-q+2 c z} - \frac{2 c}{q} \frac{z^m}{b+q+2 c z}$
- **Rule: If** $b^2 - 4 a c \neq 0 \land m, n \in \mathbb{Z} \bigwedge 0 < m < n \bigwedge \text{CoprimeQ}[m+1, n] \bigwedge \neg \left(\frac{n}{2} \in \mathbb{Z} \bigwedge b^2 - 4 a c < 0 \bigwedge a c > 0\right)$**, let** $q = \sqrt{b^2 - 4 a c}$**, then**

$$\int \frac{x^m}{a + b x^n + c x^{2 n}} \, dx \longrightarrow \frac{2 c}{q} \int \frac{x^m}{b - q + 2 c x^n} \, dx - \frac{2 c}{q} \int \frac{x^m}{b + q + 2 c x^n} \, dx$$

- **Program code:**

```
Int[x_^m_./(a_+b_.*x_^n_+c_.*x_^j_),x_Symbol] :=
  Module[{q=Rt[b^2-4*a*c,2]},
  Dist[2*c/q,Int[x^m/(b-q+2*c*x^n),x]] -
  Dist[2*c/q,Int[x^m/(b+q+2*c*x^n),x]]] /;
FreeQ[{a,b,c},x] && NonzeroQ[b^2-4*a*c] && ZeroQ[j-2*n] && IntegersQ[m,n] && 0<m<n &&
  CoprimeQ[m+1,n] && Not[IntegerQ[n/2] && NegativeQ[b^2-4*a*c] && PosQ[a*c]]
```

- **Derivation: Algebraic expansion**
- **Basis: Let** $q = \sqrt{b^2 - 4 a c}$**, then** $\frac{z^m}{a+b z+c z^2} = \left(1 - \frac{b}{q}\right) \frac{z^{m-1}}{b-q+2 c z} + \left(1 + \frac{b}{q}\right) \frac{z^{m-1}}{b+q+2 c z}$
- **Rule: If** $b^2 - 4 a c \neq 0 \land m, n \in \mathbb{Z} \bigwedge n < m < 2 n \bigwedge \text{CoprimeQ}[m+1, n] \bigwedge \neg \left(\frac{n}{2} \in \mathbb{Z} \bigwedge b^2 - 4 a c < 0 \bigwedge a c > 0\right)$**, let** $q = \sqrt{b^2 - 4 a c}$**, then**

$$\int \frac{x^m}{a + b x^n + c x^{2 n}} \, dx \longrightarrow \left(1 - \frac{b}{q}\right) \int \frac{x^{m-n}}{b - q + 2 c x^n} \, dx + \left(1 + \frac{b}{q}\right) \int \frac{x^{m-n}}{b + q + 2 c x^n} \, dx$$

- **Program code:**

```
Int[x_^m_/(a_+b_.*x_^n_+c_.*x_^j_),x_Symbol] :=
  Module[{q=Rt[b^2-4*a*c,2]},
  Dist[1-b/q,Int[x^(m-n)/(b-q+2*c*x^n),x]] +
  Dist[1+b/q,Int[x^(m-n)/(b+q+2*c*x^n),x]]] /;
FreeQ[{a,b,c},x] && NonzeroQ[b^2-4*a*c] && ZeroQ[j-2*n] && IntegersQ[m,n] && n<m<2*n &&
  CoprimeQ[m+1,n] && Not[IntegerQ[n/2] && NegativeQ[b^2-4*a*c] && PosQ[a*c]]
```

$$\int \frac{d + e\,x^n}{a + b\,x^n + c\,x^{2\,n}}\, dx$$

- **Derivation: Algebraic expansion**

- **Basis:** Let $q = \sqrt{a\,c}$ and $r = \sqrt{2\,c\,q - b\,c}$, then $\frac{d+e\,z^2}{a+b\,z^2+c\,z^4} = \frac{c}{2\,q\,r}\,\frac{d\,r-(c\,d-e\,q)\,z}{q-r\,z+c\,z^2} + \frac{c}{2\,q\,r}\,\frac{d\,r+(c\,d-e\,q)\,z}{q+r\,z+c\,z^2}$

- **Rule:** If $b^2 - 4\,a\,c < 0 \bigwedge \frac{n}{2} \in \mathbb{Z} \bigwedge n > 2 \bigwedge a\,c > 0$, let $q = \sqrt{a\,c}$, if $2\,c\,q - b\,c > 0$, let $r = \sqrt{2\,c\,q - b\,c}$, then

$$\int \frac{d + e\,x^n}{a + b\,x^n + c\,x^{2\,n}}\, dx \,\longrightarrow\, \frac{c}{2\,q\,r} \int \frac{d\,r - (c\,d - e\,q)\,x^{n/2}}{q - r\,x^{n/2} + c\,x^n}\, dx + \frac{c}{2\,q\,r} \int \frac{d\,r + (c\,d - e\,q)\,x^{n/2}}{q + r\,x^{n/2} + c\,x^n}\, dx$$

- **Program code:**

```
Int[(d_.+e_.*x_^n_)/(a_+b_.*x_^n_+c_.*x_^j_),x_Symbol] :=
  Module[{q=Rt[a*c,2]},
  Module[{r=Rt[2*c*q-b*c,2]},
  Dist[c/(2*q*r),Int[(d*r-(c*d-e*q)*x^(n/2))/(q-r*x^(n/2)+c*x^n),x]] +
  Dist[c/(2*q*r),Int[(d*r+(c*d-e*q)*x^(n/2))/(q+r*x^(n/2)+c*x^n),x]]] /;
 Not[NegativeQ[2*c*q-b*c]]] /;
FreeQ[{a,b,c,d,e},x] && NegativeQ[b^2-4*a*c] && ZeroQ[j-2*n] && IntegerQ[n/2] && n>2 && PosQ[a*c]
```

- **Reference:** G&R 2.161.1a & G&R 2.161.3

- **Derivation: Algebraic expansion**

- **Basis:** Let $q = \sqrt{b^2 - 4\,a\,c}$, then $\frac{d+e\,z}{a+b\,z+c\,z^2} = \left(e + \frac{2\,c\,d-b\,e}{q}\right) \frac{1}{b-q+2\,c\,z} + \left(e - \frac{2\,c\,d-b\,e}{q}\right) \frac{1}{b+q+2\,c\,z}$

- **Rule:** If $b^2 - 4\,a\,c \neq 0 \bigwedge n \in \mathbb{Z} \bigwedge n > 1 \bigwedge \neg\left(\frac{n}{2} \in \mathbb{Z} \bigwedge b^2 - 4\,a\,c < 0 \bigwedge a\,c > 0\right)$, let $q = \sqrt{b^2 - 4\,a\,c}$, then

$$\int \frac{d + e\,x^n}{a + b\,x^n + c\,x^{2\,n}}\, dx \,\longrightarrow\, \left(e + \frac{2\,c\,d - b\,e}{q}\right) \int \frac{1}{b - q + 2\,c\,x^n}\, dx + \left(e - \frac{2\,c\,d - b\,e}{q}\right) \int \frac{1}{b + q + 2\,c\,x^n}\, dx$$

- **Program code:**

```
Int[(d_.+e_.*x_^n_)/(a_+b_.*x_^n_+c_.*x_^j_),x_Symbol] :=
  Module[{q=Rt[b^2-4*a*c,2]},
  Dist[(e+(2*c*d-b*e)/q),Int[1/(b-q+2*c*x^n),x]] +
  Dist[(e-(2*c*d-b*e)/q),Int[1/(b+q+2*c*x^n),x]]] /;
FreeQ[{a,b,c,d,e},x] && NonzeroQ[b^2-4*a*c] && ZeroQ[j-2*n] && IntegerQ[n] && n>1 &&
  Not[IntegerQ[n/2] && NegativeQ[b^2-4*a*c] && PosQ[a*c]]
```

$$\int \frac{x^m (d + e x^n)}{a + b x^n + c x^{2n}} \, dx$$

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{d+e x^n}{x (a+b x^n+c x^{2n})} = \frac{d}{a x} - \frac{1}{a} \frac{x^{n-1} (b d-a e+c d x^n)}{a+b x^n+c x^{2n}}$
- **Note: Although resulting integrand appears more complicated than the original one, it is easily integrated by the substitution $u = x^n$.**
- **Rule: If $b^2 - 4 a c \neq 0$, then**

$$\int \frac{d + e x^n}{x \left(a + b x^n + c x^{2n}\right)} \, dx \longrightarrow \frac{d \operatorname{Log}[x]}{a} - \frac{1}{a} \int \frac{x^{n-1} (b d - a e + c d x^n)}{a + b x^n + c x^{2n}} \, dx$$

- **Program code:**

```
Int[(d_+e_.*x_^n_)/(x_*(a_+b_.*x_^n_+c_.*x_^j_)),x_Symbol] :=
  d*Log[x]/a - Dist[1/a,Int[x^(n-1)*(b*d-a*e+c*d*x^n)/(a+b*x^n+c*x^(2*n)),x]] /;
FreeQ[{a,b,c,d,e,n},x] && NonzeroQ[b^2-4*a*c] && ZeroQ[j-2*n]
```

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{x^m (d+e x^n)}{a+b x^n+c x^{2n}} = \frac{d x^m}{a} - \frac{1}{a} \frac{x^{m+n} (b d-a e+c d x^n)}{a+b x^n+c x^{2n}}$
- **Note: Resulting integrand has the same form as the original one so recursion can occur.**
- **Rule: If $b^2 - 4 a c \neq 0 \land m, n \in \mathbb{Z} \land n > 0 \land m < -1$, then**

$$\int \frac{x^m (d + e x^n)}{a + b x^n + c x^{2n}} \, dx \longrightarrow \frac{d x^{m+1}}{a (m+1)} - \frac{1}{a} \int \frac{x^{m+n} (b d - a e + c d x^n)}{a + b x^n + c x^{2n}} \, dx$$

- **Program code:**

```
Int[x_^m_*(d_+e_.*x_^n_)/(a_+b_.*x_^n_+c_.*x_^j_),x_Symbol] :=
  d*x^(m+1)/(a*(m+1)) -
  Dist[1/a,Int[x^(m+n)*(b*d-a*e+c*d*x^n)/(a+b*x^n+c*x^(2*n)),x]] /;
FreeQ[{a,b,c,d,e},x] && NonzeroQ[b^2-4*a*c] && ZeroQ[j-2*n] && IntegersQ[m,n] && n>0 && m<-1
```

- **Derivation: Algebraic expansion**

- **Basis:** $\frac{x^m\,(d+e\,x^n)}{a+b\,x^n+c\,x^{2\,n}}=\frac{e\,x^{m-n}}{c}-\frac{1}{c}\,\frac{x^{m-n}\,(a\,e+(b\,e-c\,d)\,x^n)}{a+b\,x^n+c\,x^{2\,n}}$

- **Note: Resulting integrand has the same form as the original one so recursion can occur.**

- **Rule: If** $b^2-4\,a\,c\neq 0\,\land\,m,\,n\in\mathbb{Z}\,\land\,0<n\leq m$, **then**

$$\int\frac{x^m\,(d+e\,x^n)}{a+b\,x^n+c\,x^{2\,n}}\,dx\,\longrightarrow\,\frac{e\,x^{m-n+1}}{c\,(m-n+1)}-\frac{1}{c}\int\frac{x^{m-n}\,(a\,e+(b\,e-c\,d)\,x^n)}{a+b\,x^n+c\,x^{2\,n}}\,dx$$

- **Program code:**

```
Int[x_^m_.*(d_+e_.*x_^n_)/(a_+b_.*x_^n_+c_.*x_^j_),x_Symbol] :=
  e*x^(m-n+1)/(c*(m-n+1)) -
  Dist[1/c,Int[x^(m-n)*(a*e+(b*e-c*d)*x^n)/(a+b*x^n+c*x^(2*n)),x]] /;
FreeQ[{a,b,c,d,e},x] && NonzeroQ[b^2-4*a*c] && ZeroQ[j-2*n] && IntegersQ[m,n] && 0<n≤m
```

- **Derivation: Algebraic expansion**

- **Basis: Let** $q=\sqrt{a\,c}$ **and** $r=\sqrt{2\,c\,q-b\,c}$, **then** $\frac{d+e\,z^2}{a+b\,z^2+c\,z^4}=\frac{c}{2\,q\,r}\,\frac{d\,r-(c\,d-e\,q)\,z}{q-r\,z+c\,z^2}+\frac{c}{2\,q\,r}\,\frac{d\,r+(c\,d-e\,q)\,z}{q+r\,z+c\,z^2}$

- **Rule: If** $b^2-4\,a\,c<0\,\land\,m,\,\frac{n}{2}\in\mathbb{Z}\,\bigwedge\,0<m<n\,\bigwedge\,\text{CoprimeQ}[m+1,\,n]\,\bigwedge\,a\,c>0$, **let** $q=\sqrt{a\,c}$, **if** $2\,c\,q-b\,c>0$, **let** $r=\sqrt{2\,c\,q-b\,c}$, **then**

$$\int\frac{x^m\,(d+e\,x^n)}{a+b\,x^n+c\,x^{2\,n}}\,dx\,\longrightarrow\,\frac{c}{2\,q\,r}\int\frac{x^m\left(d\,r-(c\,d-e\,q)\,x^{n/2}\right)}{q-r\,x^{n/2}+c\,x^n}\,dx+\frac{c}{2\,q\,r}\int\frac{x^m\left(d\,r+(c\,d-e\,q)\,x^{n/2}\right)}{q+r\,x^{n/2}+c\,x^n}\,dx$$

- **Program code:**

```
Int[x_^m_*(d_+e_.*x_^n_)/(a_+b_.*x_^n_+c_.*x_^j_),x_Symbol] :=
  Module[{q=Rt[a*c,2]},
  Module[{r=Rt[2*c*q-b*c,2]},
  Dist[c/(2*q*r),Int[x^m*(d*r-(c*d-e*q)*x^(n/2))/(q-r*x^(n/2)+c*x^n),x]] +
  Dist[c/(2*q*r),Int[x^m*(d*r+(c*d-e*q)*x^(n/2))/(q+r*x^(n/2)+c*x^n),x]]] /;
 Not[NegativeQ[2*c*q-b*c]]] /;
FreeQ[{a,b,c,d,e},x] && NegativeQ[b^2-4*a*c] && ZeroQ[j-2*n] && IntegersQ[m,n/2] &&
  0<m<n && CoprimeQ[m+1,n] && PosQ[a*c]
```

- **Derivation: Algebraic expansion**
- **Basis: Let** $q=\sqrt{b^2-4\,a\,c}$ **, then** $\frac{z^m\,(d+e\,z)}{a+b\,z+c\,z^2}=\left(e+\frac{2\,c\,d-b\,e}{q}\right)\frac{z^m}{b-q+2\,c\,z}+\left(e-\frac{2\,c\,d-b\,e}{q}\right)\frac{z^m}{b+q+2\,c\,z}$
- **Rule: If** $b^2-4\,a\,c\neq 0\,\wedge\,m,n\in\mathbb{Z}\,\bigwedge\,0<m<n\,\bigwedge\,\text{CoprimeQ}[m+1,n]\,\bigwedge\,\neg\left(\frac{n}{2}\in\mathbb{Z}\,\bigwedge\,b^2-4\,a\,c<0\,\bigwedge\,a\,c>0\right)$**, let** $q=\sqrt{b^2-4\,a\,c}$ **, then**

$$\int\frac{x^m\,(d+e\,x^n)}{a+b\,x^n+c\,x^{2\,n}}\,dx\,\longrightarrow\,\left(e+\frac{2\,c\,d-b\,e}{q}\right)\int\frac{x^m}{b-q+2\,c\,x^n}\,dx+\left(e-\frac{2\,c\,d-b\,e}{q}\right)\int\frac{x^m}{b+q+2\,c\,x^n}\,dx$$

- **Program code:**

```
Int[x_^m_.*(d_+e_.*x_^n_)/(a_+b_.*x_^n_+c_.*x_^j_),x_Symbol] :=
  Module[{q=Rt[b^2-4*a*c,2]},
  Dist[(e+(2*c*d-b*e)/q),Int[x^m/(b-q+2*c*x^n),x]] +
  Dist[(e-(2*c*d-b*e)/q),Int[x^m/(b+q+2*c*x^n),x]]] /;
FreeQ[{a,b,c,d,e},x] && NonzeroQ[b^2-4*a*c] && ZeroQ[j-2*n] && IntegersQ[m,n] &&
  0<m<n && CoprimeQ[m+1,n] && Not[IntegerQ[n/2] && NegativeQ[b^2-4*a*c] && PosQ[a*c]]
```

$$\int \left(a + b x^n + c x^{2n}\right)^p \, dx$$

- **Derivation: Algebraic simplification**
- **Basis: If** $b^2 - 4ac = 0$, **then** $a + b z^n + c z^{2n} = \frac{1}{c}\left(\frac{b}{2} + c z^n\right)^2$
- **Rule: If** $n, p \in \mathbb{Z} \land n > 1 \land p < 0 \land b^2 - 4ac = 0$, **then**

$$\int \left(a + b x^n + c x^{2n}\right)^p \, dx \longrightarrow \frac{1}{c^p} \int \left(\frac{b}{2} + c x^n\right)^{2p} \, dx$$

- **Program code:**

```
Int[(a_+b_.*x_^n_+c_.*x_^j_)^p_,x_Symbol] :=
  Dist[1/c^p,Int[(b/2+c*x^n)^(2*p),x]] /;
FreeQ[{a,b,c},x] && ZeroQ[j-2*n] && IntegersQ[n,p] && n>1 && p<0 && ZeroQ[b^2-4*a*c]
```

- **Reference: G&R 2.161.5**
- **Note: G&R 2.161.4 is a special case of G&R 2.161.5.**
- **Note: Previously undiscovered rule?**
- **Rule: If** $n \in \mathbb{Z} \land n > 1 \land p < -1 \land b^2 - 4ac \neq 0$, **then**

$$\int \left(a + b x^n + c x^{2n}\right)^p \, dx \longrightarrow -\frac{x \left(b^2 - 2ac + bc x^n\right) \left(a + b x^n + c x^{2n}\right)^{p+1}}{a n (p+1) \left(b^2 - 4ac\right)} +$$
$$\frac{1}{a n (p+1) \left(b^2 - 4ac\right)} \int \left(b^2 - 2ac + n (p+1) \left(b^2 - 4ac\right) + bc (n (2p+3) + 1) x^n\right) \left(a + b x^n + c x^{2n}\right)^{p+1} \, dx$$

- **Program code:**

```
Int[(a_+b_.*x_^n_+c_.*x_^j_)^p_,x_Symbol] :=
  -x*(b^2-2*a*c+b*c*x^n)*(a+b*x^n+c*x^j)^(p+1)/(a*n*(p+1)*(b^2-4*a*c)) +
  Dist[1/(a*n*(p+1)*(b^2-4*a*c)),
    Int[(b^2-2*a*c+n*(p+1)*(b^2-4*a*c)+b*c*(n*(2*p+3)+1)*x^n)*(a+b*x^n+c*x^(2*n))^(p+1),x]] /;
FreeQ[{a,b,c},x] && ZeroQ[j-2*n] && IntegerQ[n] && n>1 && RationalQ[p] && p<-1 &&
NonzeroQ[b^2-4*a*c]
```

$$\int x^m \left(a + b\, x^n + c\, x^{2\,n}\right)^p \, dx$$

- **Derivation: Algebraic simplification**
- **Basis: If** $b^2 - 4\,a\,c = 0$, **then** $a + b\, z^n + c\, z^{2\,n} = \frac{1}{c} \left(\frac{b}{2} + c\, z^n\right)^2$
- **Rule: If** $n,\ p \in \mathbb{Z} \bigwedge n > 1 \bigwedge p < 0 \bigwedge b^2 - 4\,a\,c = 0$, **then**

$$\int x^m \left(a + b\, x^n + c\, x^{2\,n}\right)^p dx \longrightarrow \frac{1}{c^p} \int x^m \left(\frac{b}{2} + c\, x^n\right)^{2\,p} dx$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*x_^n_+c_.*x_^j_)^p_,x_Symbol] :=
  Dist[1/c^p,Int[x^m*(b/2+c*x^n)^(2*p),x]] /;
FreeQ[{a,b,c},x] && ZeroQ[j-2*n] && IntegersQ[m,n,p] && n>1 && p<0 && ZeroQ[b^2-4*a*c]
```

- **Derivation: Trinomial recurrence 2 with** $d = 0$ **and** $e = 1$
- **Rule: If** $b^2 - 4\,a\,c \neq 0 \bigwedge m,\ n \in \mathbb{Z} \bigwedge 0 < n \leq m$, **then**

$$\int x^m \left(a + b\, x^n + c\, x^{2\,n}\right)^p dx \longrightarrow \frac{x^{m-2\,n+1} \left(a + b\, x^n + c\, x^{2\,n}\right)^{p+1}}{c\,(2\,n\,p + m + 1)} -$$
$$\frac{1}{c\,(2\,n\,p + m + 1)} \int x^{m-2\,n} \left(a\,(m - 2\,n + 1) + (b\,(n\,p - n + m + 1))\, x^n\right) \left(a + b\, x^n + c\, x^{2\,n}\right)^p dx$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*x_^n_.+c_.*x_^j_)^p_.,x_Symbol] :=
  x^(m-2*n+1)*(a+b*x^n+c*x^(2*n))^(p+1)/(c*(2*n*p+m+1)) -
  Dist[1/(c*(2*n*p+m+1)),
    Int[x^(m-2*n)*Sim[a*(m-2*n+1)+(b*(n*p-n+m+1))*x^n,x]*(a+b*x^n+c*x^(2*n))^p,x]] /;
FreeQ[{a,b,c,p},x] && NonzeroQ[b^2-4*a*c] && ZeroQ[j-2*n] && IntegersQ[m,n] && 0<n≤m
```

- **Reference: G&R 2.160.1 special case**

- **Rule: If** $m, n \in \mathbb{Z} \land m < -1 \land n > 0 \land m + n (p + 1) + 1 = 0 \land \neg (p \in \mathbb{Z} \land p > 0)$, **then**

$$\int x^m \left(a + b x^n + c x^{2n}\right)^p dx \longrightarrow \frac{x^{m+1} \left(a + b x^n + c x^{2n}\right)^{p+1}}{a (m + 1)} + \frac{c}{a} \int x^{m+2n} \left(a + b x^n + c x^{2n}\right)^p dx$$

- **Program code:**

```
Int[x_^m_*(a_+b_.*x_^n_.+c_.*x_^j_.)^p_,x_Symbol] :=
  x^(m+1)*(a+b*x^n+c*x^j)^(p+1)/(a*(m+1)) +
  Dist[c/a,Int[x^(m+2*n)*(a+b*x^n+c*x^j)^p,x]] /;
FreeQ[{a,b,c,p},x] && ZeroQ[j-2*n] && IntegersQ[m,n] && m<-1 && n>0 && ZeroQ[m+n*(p+1)+1] &&
Not[IntegerQ[p] && p>0]
```

- **Reference: G&R 2.160.1 special case**

- **Rule: If** $m, n \in \mathbb{Z} \land m < -1 \land n > 0 \land m + 2n (p + 1) + 1 = 0 \land \neg (p \in \mathbb{Z} \land p > 0)$, **then**

$$\int x^m \left(a + b x^n + c x^{2n}\right)^p dx \longrightarrow \frac{x^{m+1} \left(a + b x^n + c x^{2n}\right)^{p+1}}{a (m + 1)} - \frac{b}{2a} \int x^{m+n} \left(a + b x^n + c x^{2n}\right)^p dx$$

- **Program code:**

```
Int[x_^m_*(a_+b_.*x_^n_.+c_.*x_^j_.)^p_,x_Symbol] :=
  x^(m+1)*(a+b*x^n+c*x^j)^(p+1)/(a*(m+1)) -
  Dist[b/(2*a),Int[x^(m+n)*(a+b*x^n+c*x^j)^p,x]] /;
FreeQ[{a,b,c,p},x] && ZeroQ[j-2*n] && IntegersQ[m,n] && m<-1 && n>0 && ZeroQ[m+2*n*(p+1)+1] &&
Not[IntegerQ[p] && p>0]
```

- **Derivation: Integration by substitution**

- **Basis: If** $\frac{n}{m+1} \in \mathbb{Z}$, **then** $x^m f[x^n] = \frac{1}{m+1} f\left[\left(x^{m+1}\right)^{\frac{n}{m+1}}\right] \partial_x x^{m+1}$

- **Rule: If** $b^2 - 4 a c \neq 0 \land m + 1 \neq 0 \land p, \frac{n}{m+1} \in \mathbb{Z} \bigwedge \neg (m + 1 < 0) \bigwedge p < 0$, **then**

$$\int x^m \left(a + b x^n + c x^{2n}\right)^p dx \longrightarrow \frac{1}{m + 1} \text{Subst}\left[\int \left(a + b x^{\frac{n}{m+1}} + c x^{\frac{2n}{m+1}}\right)^p dx, x, x^{m+1}\right]$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*x_^n_+c_.*x_^j_)^p_,x_Symbol] :=
  Dist[1/(m+1),Subst[Int[(a+b*x^(n/(m+1))+c*x^(2*n/(m+1)))^p,x],x,x^(m+1)]] /;
FreeQ[{a,b,c,m,n},x] && NonzeroQ[b^2-4*a*c] && ZeroQ[j-2*n] && NonzeroQ[m+1] &&
  IntegersQ[p,n/(m+1)] && Not[NegativeQ[m+1]] && p<0
```

- **Derivation: Integration by substitution**

- **Basis: If** $\frac{m+1}{n} \in \mathbb{Z}$**, then** $x^m f[x^n] = \frac{1}{n} (x^n)^{\frac{m+1}{n}-1} f[x^n] \partial_x x^n$

- **Note: Requirement that** $\frac{m+1}{n} > 0$ **ensures** $\text{Log}[x]$ **rather than** $\frac{\text{Log}[x^n]}{n}$ **occurs in the antiderivative.**

- **Rule: If** $b^2 - 4ac \neq 0 \land p, \frac{m+1}{n} \in \mathbb{Z} \bigwedge n \neq 1 \bigwedge \neg (n < 0) \bigwedge p < 0 \bigwedge \left(\frac{m+1}{n} > 0 \bigvee m \notin \mathbb{Z}\right)$**, then**

$$\int x^m \left(a + b x^n + c x^{2n}\right)^p dx \longrightarrow \frac{1}{n} \text{Subst}\left[\int x^{\frac{m+1}{n}-1} \left(a + b x + c x^2\right)^p dx, x, x^n\right]$$

```
Int[x_^m_.*(a_+b_.*x_^n_+c_.*x_^j_)^p_,x_Symbol] :=
  Dist[1/n,Subst[Int[x^((m+1)/n-1)*(a+b*x+c*x^2)^p,x],x,x^n]] /;
FreeQ[{a,b,c,m,n},x] && NonzeroQ[b^2-4*a*c] && ZeroQ[j-2*n] && IntegersQ[p,(m+1)/n] &&
  Not[NegativeQ[n]] && p<0 && ((m+1)/n>0 || Not[IntegerQ[m]])
```

- **Derivation: Integration by substitution**

- **Rule: If** $b^2 - 4ac \neq 0 \land m + 1 \neq 0 \land n \neq 1 \land m, n, p \in \mathbb{Z} \bigwedge \neg \text{CoprimeQ}[m + 1, n] \bigwedge p < 0 \bigwedge \frac{m+1}{n} > 0$**, let** $g = \text{GCD}[m + 1, n]$**, then**

$$\int x^m \left(a + b x^n + c x^{2n}\right)^p dx \longrightarrow \frac{1}{g} \text{Subst}\left[\int x^{\frac{m+1}{g}-1} \left(a + b x^{\frac{n}{g}} + c x^{\frac{2n}{g}}\right)^p dx, x, x^g\right]$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*x_^n_+c_.*x_^j_)^p_,x_Symbol] :=
  Module[{g=GCD[m+1,n]},
  Dist[1/g,Subst[Int[x^((m+1)/g-1)*(a+b*x^(n/g)+c*x^(2*n/g))^p,x],x,x^g]]] /;
FreeQ[{a,b,c,m,n},x] && NonzeroQ[b^2-4*a*c] && ZeroQ[j-2*n] && NonzeroQ[m+1] &&
  IntegersQ[m,n,p] && Not[CoprimeQ[m+1,n]] && p<0 && (m+1)/n>0
```

$$\int (d + e\,x^n)\,\left(a + b\,x^n + c\,x^{2\,n}\right)^p\,dx$$

- **Derivation: Algebraic simplification**

- **Basis: If $b^2 - 4\,a\,c = 0$, then $a + b\,z^n + c\,z^{2\,n} = \frac{1}{c}\left(\frac{b}{2} + c\,z^n\right)^2$**

- **Rule: If $n, p \in \mathbb{Z} \wedge n > 1 \wedge p < 0 \wedge b^2 - 4\,a\,c = 0$, then**

$$\int (d + e\,x^n)\,\left(a + b\,x^n + c\,x^{2\,n}\right)^p\,dx \longrightarrow \frac{1}{c^p}\int (d + e\,x^n)\,\left(\frac{b}{2} + c\,x^n\right)^{2\,p}\,dx$$

- **Program code:**

```
Int[(d_.+e_.*x_^n_)*(a_+b_.*x_^n_+c_.*x_^j_)^p_,x_Symbol] :=
  Dist[1/c^p,Int[(d+e*x^n)*(b/2+c*x^n)^(2*p),x]] /;
FreeQ[{a,b,c,d,e},x] && ZeroQ[j-2*n] && IntegersQ[n,p] && n>1 && p<0 && ZeroQ[b^2-4*a*c]
```

- **Note: Previously undiscovered rule?**

- **Note: Since the resulting integrand has the same form as the original one, recursion can occur.**

- **Rule: If $n \in \mathbb{Z} \wedge n > 1 \wedge p < -1 \wedge b^2 - 4\,a\,c \neq 0$, then**

$$\int (d + e\,x^n)\,\left(a + b\,x^n + c\,x^{2\,n}\right)^p\,dx \longrightarrow$$

$$\frac{x\,\left(a\,b\,e - b^2\,d + 2\,a\,c\,d + c\,(2\,a\,e - b\,d)\,x^n\right)}{a\,n\,(p+1)\,\left(b^2 - 4\,a\,c\right)}\,\left(a + b\,x^n + c\,x^{2\,n}\right)^{p+1} - \frac{1}{a\,n\,(p+1)\,\left(b^2 - 4\,a\,c\right)}\,\cdot$$

$$\int \left(a\,b\,e - b^2\,d + 2\,a\,c\,d - d\,n\,(p+1)\,\left(b^2 - 4\,a\,c\right) + c\,(2\,a\,e - b\,d)\,(n\,(2\,p+3)+1)\,x^n\right)\,\left(a + b\,x^n + c\,x^{2\,n}\right)^{p+1}\,dx$$

- **Program code:**

```
Int[(d_.+e_.*x_^n_)*(a_+b_.*x_^n_+c_.*x_^j_)^p_,x_Symbol] :=
  x*(a*b*e-b^2*d+2*a*c*d+c*(2*a*e-b*d)*x^n)*(a+b*x^n+c*x^(2*n))^(p+1)/(a*n*(p+1)*(b^2-4*a*c)) -
  Dist[1/(a*n*(p+1)*(b^2-4*a*c)),
    Int[(a*b*e-b^2*d+2*a*c*d-d*n*(p+1)*(b^2-4*a*c)+c*(2*a*e-b*d)*(n*(2*p+3)+1)*x^n)*
      (a+b*x^n+c*x^(2*n))^(p+1),x]] /;
FreeQ[{a,b,c,d,e},x] && ZeroQ[j-2*n] && IntegerQ[n] && n>1 && RationalQ[p] && p<-1 &&
NonzeroQ[b^2-4*a*c]
```

$$\int x^m (d + e x^n) \left(a + b x^n + c x^{2n}\right)^p \, dx$$

- **Derivation: Trinomial recurrence 2**
- **Rule: If** $b^2 - 4 a c \neq 0 \land m, n \in \mathbb{Z} \land 0 < n \leq m$, **then**

$$\int x^m (d + e x^n) \left(a + b x^n + c x^{2n}\right)^p dx \longrightarrow \frac{e x^{m-n+1} \left(a + b x^n + c x^{2n}\right)^{p+1}}{c (2 n p + n + m + 1)} -$$
$$\frac{1}{c (2 n p + n + m + 1)} \int x^{m-n} (a e (m - n + 1) + (b e (n p + m + 1) - c d (2 n p + n + m + 1)) x^n) \left(a + b x^n + c x^{2n}\right)^p dx$$

- **Program code:**

```
Int[x_^m_.*(d_+e_.*x_^n_.)*(a_+b_.*x_^n_.+c_.*x_^j_)^p_.,x_Symbol] :=
  e*x^(m-n+1)*(a+b*x^n+c*x^(2*n))^(p+1)/(c*(2*n*p+n+m+1)) -
  Dist[1/(c*(2*n*p+n+m+1)),
    Int[x^(m-n)*Sim[a*e*(m-n+1)+(b*e*(n*p+m+1)-c*d*(2*n*p+n+m+1))*x^n,x]*(a+b*x^n+c*x^(2*n))^p,x]] /
FreeQ[{a,b,c,d,e,p},x] && NonzeroQ[b^2-4*a*c] && ZeroQ[j-2*n] && IntegersQ[m,n] && 0<n≤m
```

- **Derivation: Integration by substitution**
- **Basis: If** $\frac{n}{m+1} \in \mathbb{Z}$, **then** $x^m f[x^n] = \frac{1}{m+1} f\left[\left(x^{m+1}\right)^{\frac{n}{m+1}}\right] \partial_x x^{m+1}$
- **Rule: If** $b^2 - 4 a c \neq 0 \land m + 1 \neq 0 \land p, \frac{n}{m+1} \in \mathbb{Z} \bigwedge \neg (m + 1 < 0) \bigwedge p < 0$, **then**

$$\int x^m (d + e x^n) \left(a + b x^n + c x^{2n}\right)^p dx \longrightarrow \frac{1}{m+1} \text{Subst}\left[\int \left(d + e x^{\frac{n}{m+1}}\right) \left(a + b x^{\frac{n}{m+1}} + c x^{\frac{2n}{m+1}}\right)^p dx, x, x^{m+1}\right]$$

- **Program code:**

```
Int[x_^m_.*(d_+e_.*x_^n_)*(a_+b_.*x_^n_+c_.*x_^j_)^p_,x_Symbol] :=
  Dist[1/(m+1),Subst[Int[(d+e*x^(n/(m+1)))*(a+b*x^(n/(m+1))+c*x^(2*n/(m+1)))^p,x],x,x^(m+1)]] /;
FreeQ[{a,b,c,d,e,m,n},x] && NonzeroQ[b^2-4*a*c] && ZeroQ[j-2*n] && NonzeroQ[m+1] &&
  IntegersQ[p,n/(m+1)] && Not[NegativeQ[m+1]] && p<0
```

- **Derivation: Integration by substitution**
- **Basis: If** $\frac{m+1}{n} \in \mathbb{Z}$, **then** $x^m f[x^n] = \frac{1}{n} (x^n)^{\frac{m+1}{n}-1} f[x^n] \partial_x x^n$
- **Note: Requirement that** $\frac{m+1}{n} > 0$ **ensures** $\text{Log}[x]$ **rather than** $\frac{\text{Log}[x^n]}{n}$ **occurs in the antiderivative.**
- **Rule: If** $b^2 - 4 a c \neq 0 \land p,\ \frac{m+1}{n} \in \mathbb{Z} \bigwedge n \neq 1 \bigwedge \neg (n < 0) \bigwedge p < 0 \bigwedge \left(\frac{m+1}{n} > 0 \bigvee m \notin \mathbb{Z}\right)$, **then**

$$\int x^m (d + e x^n) \left(a + b x^n + c x^{2n}\right)^p dx \longrightarrow \frac{1}{n} \text{Subst}\left[\int x^{\frac{m+1}{n}-1} (d + e x) \left(a + b x + c x^2\right)^p dx,\ x,\ x^n\right]$$

- **Program code:**

```
Int[x_^m_.*(d_+e_.*x_^n_)*(a_+b_.*x_^n_+c_.*x_^j_)^p_,x_Symbol] :=
  Dist[1/n,Subst[Int[x^((m+1)/n-1)*(d+e*x)*(a+b*x+c*x^2)^p,x],x,x^n]] /;
FreeQ[{a,b,c,d,e,m,n},x] && NonzeroQ[b^2-4*a*c] && ZeroQ[j-2*n] && IntegersQ[p,(m+1)/n] &&
  Not[NegativeQ[n]] && p<0 && ((m+1)/n>0 || Not[IntegerQ[m]])
```

- **Derivation: Integration by substitution**
- **Rule: If** $b^2 - 4 a c \neq 0 \land m + 1 \neq 0 \land n \neq 1 \land m,\ n,\ p \in \mathbb{Z} \bigwedge \neg \text{CoprimeQ}[m + 1,\ n] \bigwedge p < 0 \bigwedge \frac{m+1}{n} > 0$, **let** $g = \text{GCD}[m + 1,\ n]$, **then**

$$\int x^m (d + e x^n) \left(a + b x^n + c x^{2n}\right)^p dx \longrightarrow \frac{1}{g} \text{Subst}\left[\int x^{\frac{m+1}{g}-1} \left(d + e x^{\frac{n}{g}}\right) \left(a + b x^{\frac{n}{g}} + c x^{\frac{2n}{g}}\right)^p dx,\ x,\ x^g\right]$$

- **Program code:**

```
Int[x_^m_.*(d_+e_.*x_^n_)*(a_+b_.*x_^n_+c_.*x_^j_)^p_,x_Symbol] :=
  Module[{g=GCD[m+1,n]},
  Dist[1/g,Subst[Int[x^((m+1)/g-1)*(d+e*x^(n/g))*(a+b*x^(n/g)+c*x^(2*n/g))^p,x],x,x^g]]] /;
FreeQ[{a,b,c,d,e,m,n},x] && NonzeroQ[b^2-4*a*c] && ZeroQ[j-2*n] && NonzeroQ[m+1] &&
  IntegersQ[m,n,p] && Not[CoprimeQ[m+1,n]] && p<0 && (m+1)/n>0
```

$$\int \frac{a + b\, x^n}{c + d\, x^2 + e\, x^n + f\, x^{2\,n}}\, dx$$

- **Note: Previously undiscovered rule?**

- **Rule: If** $b^2\, c - a^2\, f\, (n-1)^2 = 0 \,\wedge\, b\, e + 2\, a\, f\, (n-1) = 0 \,\wedge\, c\, d > 0$**, then**

$$\int \frac{a + b\, x^n}{c + d\, x^2 + e\, x^n + f\, x^{2\,n}}\, dx \longrightarrow \frac{a}{\sqrt{c\, d}}\, \text{ArcTan}\left[\frac{a\, (n-1)\, \sqrt{c\, d}\, x}{c\, (a\, (n-1) - b\, x^n)}\right]$$

- **Program code:**

```
Int[(a_+b_.*x_^n_)/(c_+d_.*x_^2+e_.*x_^n_+f_.*x_^j_), x_Symbol] :=
  a/Rt[c*d,2]*ArcTan[(n-1)*a*Rt[c*d,2]*x/(c*(a*(n-1)-b*x^n))] /;
FreeQ[{a,b,c,d,e,f,n},x] && ZeroQ[j-2*n] &&
ZeroQ[b^2*c-a^2*f*(n-1)^2] && ZeroQ[b*e+2*a*f*(n-1)] && PosQ[c*d]
```

- **Note: Previously undiscovered rule?**

- **Rule: If** $b^2\, c - a^2\, f\, (n-1)^2 = 0 \,\wedge\, b\, e + 2\, a\, f\, (n-1) = 0 \,\wedge\, \neg\, (c\, d > 0)$**, then**

$$\int \frac{a + b\, x^n}{c + d\, x^2 + e\, x^n + f\, x^{2\,n}}\, dx \longrightarrow \frac{a}{\sqrt{-c\, d}}\, \text{ArcTanh}\left[\frac{a\, (n-1)\, \sqrt{-c\, d}\, x}{c\, (a\, (n-1) - b\, x^n)}\right]$$

- **Program code:**

```
Int[(a_+b_.*x_^n_)/(c_+d_.*x_^2+e_.*x_^n_+f_.*x_^j_), x_Symbol] :=
  a/Rt[-c*d,2]*ArcTanh[a*(n-1)*Rt[-c*d,2]*x/(c*(a*(n-1)-b*x^n))] /;
FreeQ[{a,b,c,d,e,f,n},x] && ZeroQ[j-2*n] &&
ZeroQ[b^2*c-a^2*f*(n-1)^2] && ZeroQ[b*e+2*a*f*(n-1)] && NegQ[c*d]
```

$$\int \frac{x^m \left(a + b x^n\right)}{c + d x^{2 (m+1)} + e x^n + f x^{2 n}} \, dx$$

- **Note: Previously undiscovered rule?**

- **Rule: If** $a^2 f (m - n + 1)^2 - b^2 c (m + 1)^2 = 0 \land b e (m + 1) - 2 a f (m - n + 1) = 0 \land c d > 0$, **then**

$$\int \frac{x^m \left(a + b x^n\right)}{c + d x^{2 (m+1)} + e x^n + f x^{2 n}} \, dx \rightarrow \frac{a}{(m + 1) \sqrt{c d}} \text{ArcTan}\left[\frac{a (m - n + 1) \sqrt{c d} \; x^{m+1}}{c \left(a (m - n + 1) + b (m + 1) x^n\right)}\right]$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*x_^n_.)/(c_+d_.*x_^k_.+e_.*x_^n_.+f_.*x_^j_), x_Symbol] :=
  a*ArcTan[a*(m-n+1)*Rt[c*d,2]*x^(m+1)/(c*(a*(m-n+1)+b*(m+1)*x^n))]/((m+1)*Rt[c*d,2]) /;
FreeQ[{a,b,c,d,e,f,m,n},x] && ZeroQ[j-2*n] && ZeroQ[k-2*(m+1)] &&
ZeroQ[a^2*f*(m-n+1)^2-b^2*c*(m+1)^2] && ZeroQ[b*e*(m+1)-2*a*f*(m-n+1)] && PosQ[c*d]
```

- **Note: Previously undiscovered rule?**

- **Rule: If** $a^2 f (m - n + 1)^2 - b^2 c (m + 1)^2 = 0 \land b e (m + 1) - 2 a f (m - n + 1) = 0 \land \neg (c d > 0)$, **then**

$$\int \frac{x^m \left(a + b x^n\right)}{c + d x^{2 (m+1)} + e x^n + f x^{2 n}} \, dx \rightarrow \frac{a}{(m + 1) \sqrt{-c d}} \text{ArcTanh}\left[\frac{a (m - n + 1) \sqrt{-c d} \; x^{m+1}}{c \left(a (m - n + 1) + b (m + 1) x^n\right)}\right]$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*x_^n_.)/(c_+d_.*x_^k_.+e_.*x_^n_.+f_.*x_^j_), x_Symbol] :=
  a*ArcTanh[a*(m-n+1)*Rt[-c*d,2]*x^(m+1)/(c*(a*(m-n+1)+b*(m+1)*x^n))]/((m+1)*Rt[-c*d,2]) /;
FreeQ[{a,b,c,d,e,f,m,n},x] && ZeroQ[j-2*n] && ZeroQ[k-2*(m+1)] &&
ZeroQ[a^2*f*(m-n+1)^2-b^2*c*(m+1)^2] && ZeroQ[b*e*(m+1)-2*a*f*(m-n+1)] && NegQ[c*d]
```

$$\int \frac{d + e\,x + f\,x^2 + g\,x^3}{a + b\,x + c\,x^2 + b\,x^3 + a\,x^4}\,dx$$

- **Derivation: Algebraic expansion**
- **Basis: If** $q = \sqrt{8\,a^2 + b^2 - 4\,a\,c}$, **then** $a + b\,x + c\,x^2 + b\,x^3 + a\,x^4 = a\left(1 + \frac{(b-q)\,x}{2\,a} + x^2\right)\left(1 + \frac{(b+q)\,x}{2\,a} + x^2\right)$
- **Basis: If** $q = \sqrt{8\,a^2 + b^2 - 4\,a\,c}$, **then** $\frac{d+e\,x+f\,x^2+g\,x^3}{a+b\,x+c\,x^2+b\,x^3+a\,x^4} = \frac{b\,d-2\,a\,e+2\,a\,g+d\,q+(2\,a\,d-2\,a\,f+b\,g+g\,q)\,x}{q\,\left(2\,a+(b+q)\,x+2\,a\,x^2\right)} - \frac{b\,d-2\,a\,e+2\,a\,g-d\,q+(2\,a\,d-2\,a\,f+b\,g-g\,q)\,x}{q\,\left(2\,a+(b-q)\,x+2\,a\,x^2\right)}$
- **Rule: If** $8\,a^2 + b^2 - 4\,a\,c > 0$, **then**

$$\int \frac{d + e\,x + f\,x^2 + g\,x^3}{a + b\,x + c\,x^2 + b\,x^3 + a\,x^4}\,dx \longrightarrow \frac{1}{q}\int \frac{b\,d - 2\,a\,e + 2\,a\,g + d\,q + (2\,a\,d - 2\,a\,f + b\,g + g\,q)\,x}{2\,a + (b + q)\,x + 2\,a\,x^2}\,dx - \frac{1}{q}\int \frac{b\,d - 2\,a\,e + 2\,a\,g - d\,q + (2\,a\,d - 2\,a\,f + b\,g - g\,q)\,x}{2\,a + (b - q)\,x + 2\,a\,x^2}\,dx$$

- **Program code:**

```
Int[(d_.+e_.*x_+f_.*x_^2+g_.*x_^3)/(a_+b_.*x_+c_.*x_^2+b_.*x_^3+a_.*x_^4),x_Symbol] :=
  Module[{q=Sqrt[8*a^2+b^2-4*a*c]},
  Dist[1/q,Int[(b*d-2*a*e+2*a*g+d*q+(2*a*d-2*a*f+b*g+g*q)*x)/(2*a+(b+q)*x+2*a*x^2),x]] -
  Dist[1/q,Int[(b*d-2*a*e+2*a*g-d*q+(2*a*d-2*a*f+b*g-g*q)*x)/(2*a+(b-q)*x+2*a*x^2),x]]] /;
FreeQ[{a,b,c,d,e,f,g},x] && PosQ[8*a^2+b^2-4*a*c]
```

- **Derivation: Algebraic expansion**
- **Basis: If** $q = \sqrt{8\,a^2 + b^2 - 4\,a\,c}$, **then** $\frac{d+e\,x+g\,x^3}{a+b\,x+c\,x^2+b\,x^3+a\,x^4} = \frac{b\,d-2\,a\,e+2\,a\,g+d\,q+(2\,a\,d+b\,g+g\,q)\,x}{q\,\left(2\,a+(b+q)\,x+2\,a\,x^2\right)} - \frac{b\,d-2\,a\,e+2\,a\,g-d\,q+(2\,a\,d+b\,g-g\,q)\,x}{q\,\left(2\,a+(b-q)\,x+2\,a\,x^2\right)}$
- **Rule: If** $8\,a^2 + b^2 - 4\,a\,c > 0$, **then**

$$\int \frac{d + e\,x + g\,x^3}{a + b\,x + c\,x^2 + b\,x^3 + a\,x^4}\,dx \longrightarrow \frac{1}{q}\int \frac{b\,d - 2\,a\,e + 2\,a\,g + d\,q + (2\,a\,d + b\,g + g\,q)\,x}{2\,a + (b + q)\,x + 2\,a\,x^2}\,dx - \frac{1}{q}\int \frac{b\,d - 2\,a\,e + 2\,a\,g - d\,q + (2\,a\,d + b\,g - g\,q)\,x}{2\,a + (b - q)\,x + 2\,a\,x^2}\,dx$$

- **Program code:**

```
Int[(d_.+e_.*x_+g_.*x_^3)/(a_+b_.*x_+c_.*x_^2+b_.*x_^3+a_.*x_^4),x_Symbol] :=
  Module[{q=Sqrt[8*a^2+b^2-4*a*c]},
  Dist[1/q,Int[(b*d-2*a*e+2*a*g+d*q+(2*a*d+b*g+g*q)*x)/(2*a+(b+q)*x+2*a*x^2),x]] -
  Dist[1/q,Int[(b*d-2*a*e+2*a*g-d*q+(2*a*d+b*g-g*q)*x)/(2*a+(b-q)*x+2*a*x^2),x]]] /;
FreeQ[{a,b,c,d,e,g},x] && PosQ[8*a^2+b^2-4*a*c]
```

$$\int \frac{\text{Pm}[x]}{\text{Qn}[x]^p}\, dx$$

- **Reference: G&R 2.104**

- **Note: Equivalent to the Ostrogradskiy-Hermite method but without the need to solve a system of linear equations.**

- **Note: Finds one term of the rational part of the antiderivative, thereby reducing the degree of the polynomial in the numerator of the integrand.**

- **Note: Requirement that $m < 2n - 1$ ensures new term is a proper fraction.**

- **Rule: If $p > 1 \wedge 1 < n \le m+1 \wedge m+1-n\,p < 0$, let $c = \frac{\text{pm}}{\text{qn}\,(m+1-n\,p)}$, then**

$$\int \frac{\text{Pm}[x]}{\text{Qn}[x]^p}\, dx \longrightarrow \frac{c\,x^{m-n+1}}{\text{Qn}[x]^{p-1}} + \int \frac{\text{Pm}[x] - c\,x^{m-n}\,((m-n+1)\,\text{Qn}[x] + (1-p)\,x\,\partial_x \text{Qn}[x])}{\text{Qn}[x]^p}\, dx$$

- **Program code:**

```
If[ShowSteps,

Int[u_*v_^p_,x_Symbol] :=
  Module[{m=Exponent[u,x],n=Exponent[v,x]},
  Module[{c=Coefficient[u,x,m]/(Coefficient[v,x,n]*(m+1+n*p)),w},
  w=Apart[u-c*x^(m-n)*((m-n+1)*v+(p+1)*x*D[v,x]),x];
  If[ZeroQ[w],
    ShowStep["
If p>1, 1<n<=m+1, and m+1-n*p<0, let c=pm/(qn*(m+1-n*p)), then if (Pm[x]-c*x^(m-n)*((m-n+1)*Qn[x]+(1-
      "Int[Pm[x]/Qn[x]^p,x]", "c*x^(m-n+1)/Qn[x]^(p-1)",
      Hold[c*x^(m-n+1)*v^(p+1)]],
  ShowStep["If p>1, 1<n<=m+1, and m+1-n*p<0, let c=pm/(qn*(m+1-n*p)), then",
    "Int[Pm[x]/Qn[x]^p,x]",
    "c*x^(m-n+1)/Qn[x]^(p-1)+Int[(Pm[x]-c*x^(m-n)*((m-n+1)*Qn[x]+(1-p)*x*D[Qn[x],x]))/Qn[x]^p,x]",
    Hold[c*x^(m-n+1)*v^(p+1) + Int[w*v^p,x]]]]] /;
 m+1>=n>1 && m+n*p<-1 && FalseQ[DerivativeDivides[v,u,x]]] /;
SimplifyFlag && RationalQ[p] && p<-1 && PolynomialQ[u,x] && PolynomialQ[v,x] && SumQ[v] &&
Not[MonomialQ[u,x] && BinomialQ[v,x]] &&
Not[ZeroQ[Coefficient[u,x,0]] && ZeroQ[Coefficient[v,x,0]]],

Int[u_*v_^p_,x_Symbol] :=
  Module[{m=Exponent[u,x],n=Exponent[v,x]},
  Module[{c=Coefficient[u,x,m]/(Coefficient[v,x,n]*(m+1+n*p)),w},
  c=Coefficient[u,x,m]/(Coefficient[v,x,n]*(m+1+n*p));
  w=Apart[u-c*x^(m-n)*((m-n+1)*v+(p+1)*x*D[v,x]),x];
  If[ZeroQ[w],
    c*x^(m-n+1)*v^(p+1),
  c*x^(m-n+1)*v^(p+1) + Int[w*v^p,x]]] /;
 m+1>=n>1 && m+n*p<-1 && FalseQ[DerivativeDivides[v,u,x]]] /;
RationalQ[p] && p<-1 && PolynomialQ[u,x] && PolynomialQ[v,x] && SumQ[v] &&
Not[MonomialQ[u,x] && BinomialQ[v,x]] &&
Not[ZeroQ[Coefficient[u,x,0]] && ZeroQ[Coefficient[v,x,0]]]]
```

$$\int u + v \, dx$$

- **Reference: G&R 2.02.5**
- **Rule:**

$$\int f'[u] \, g[v] \, \partial_x u + f[u] \, g'[v] \, \partial_x v \, dx \longrightarrow f[u] \, g[v]$$

- **Program code:**

```
Int[f_'[u_]*g_[v_]*w_. + f_[u_]*g_'[v_]*t_.,x_Symbol] :=
  f[u]*g[v] /;
FreeQ[{f,g},x] && ZeroQ[w-D[u,x]] && ZeroQ[t-D[v,x]]
```

- **Reference: G&R 2.02.2, CRC 2,4**
- **Rule:**

$$\int a + b\,u + c\,v + \cdots \, dx \longrightarrow a\,x + b \int u \, dx + c \int v \, dx + \cdots$$

- **Program code:**

```
If[ShowSteps,

Int[u_,x_Symbol] :=
  If[SplitFreeTerms[u,x][[1]]===0,
    ShowStep["","Int[a*u+b*v+⋯,x]","a*Int[u,x]+b*Int[v,x]+⋯",Hold[
    SplitFreeIntegrate[u,x]]],
  ShowStep["","Int[a+b*u+c*v+⋯,x]","a*x+b*Int[u,x]+c*Int[v,x]+⋯",Hold[
  SplitFreeIntegrate[u,x]]]] /;
SimplifyFlag && SumQ[u],

Int[u_,x_Symbol] :=
  SplitFreeIntegrate[u,x] /;
SumQ[u]]
```

```
SplitFreeIntegrate[u_,x_Symbol] :=
  If[SumQ[u],
    Map[Function[SplitFreeIntegrate[#,x]],u],
  If[FreeQ[u,x],
    u*x,
  If[MatchQ[u,c_*(a_+b_.*x) /; FreeQ[{a,b,c},x]],
    Int[u,x],
  Module[{lst=SplitFreeFactors[u,x]},
  Dist[lst[[1]], Int[lst[[2]],x]]]]]]
```