$$\int u\,(c\,F[a+b\,x]^n)^m\,dx$$

- **Derivation: Piecewise constant extraction**
- **Basis:** $\partial_z \frac{\sqrt{c\,F[z]^n}}{F[z]^{n/2}} = 0$
- **Rule: If** $\frac{n}{2},\ m-\frac{1}{2} \in \mathbb{Z} \bigwedge m > 0$, **then**

$$\int u\,(c\,F[a+b\,x]^n)^m\,dx \longrightarrow c^{m-\frac{1}{2}}\,\frac{\sqrt{c\,F[a+b\,x]^n}}{F[a+b\,x]^{n/2}}\int u\,F[a+b\,x]^{m\,n}\,dx$$

- **Program code:**

```
Int[u_.*(c_.*F_[a_.+b_.*x_]^n_)^m_,x_Symbol] :=
  Dist[c^(m-1/2)*Sqrt[c*F[a+b*x]^n]/F[a+b*x]^(n/2),Int[u*F[a+b*x]^(m*n),x]] /;
FreeQ[{a,b,c},x] && IntegerQ[n/2] && IntegerQ[m-1/2] && m>0 &&
  MemberQ[{Sin,Cos,Tan,Cot,Sinh,Cosh,Tanh,Coth},F]
```

- **Derivation: Piecewise constant extraction**
- **Basis:** $\partial_z \frac{F[z]^{n/2}}{\sqrt{c\,F[z]^n}} = 0$
- **Rule: If** $\frac{n}{2},\ m-\frac{1}{2} \in \mathbb{Z} \bigwedge m < 0$, **then**

$$\int u\,(c\,F[a+b\,x]^n)^m\,dx \longrightarrow c^{m+\frac{1}{2}}\,\frac{F[a+b\,x]^{n/2}}{\sqrt{c\,F[a+b\,x]^n}}\int u\,F[a+b\,x]^{m\,n}\,dx$$

- **Program code:**

```
Int[u_.*(c_.*F_[a_.+b_.*x_]^n_)^m_,x_Symbol] :=
  Dist[c^(m+1/2)*F[a+b*x]^(n/2)/Sqrt[c*F[a+b*x]^n],Int[u*F[a+b*x]^(m*n),x]] /;
FreeQ[{a,b,c},x] && IntegerQ[n/2] && IntegerQ[m-1/2] && m<0 &&
  MemberQ[{Sin,Cos,Tan,Cot,Sinh,Cosh,Tanh,Coth},F]
```

$$\int f[\text{Sin}[u]]\, \partial_x \text{Sin}[u]\, dx$$

- **Derivation: Integration by substitution**
- **Basis:** $f[\text{Sin}[z]]\, \text{Cos}[z] = f[\text{Sin}[z]]\, \partial_z \text{Sin}[z]$
- **Rule:**

$$\int f[\text{Sin}[a+b\,x]]\, \text{Cos}[a+b\,x]\, dx \longrightarrow \frac{1}{b}\, \text{Subst}\left[\int f[x]\, dx,\, x,\, \text{Sin}[a+b\,x]\right]$$

- **Program code:**

```
Int[u_*Cos[c_.*(a_.+b_.*x_)],x_Symbol] :=
  Dist[1/(b*c),Subst[Int[Regularize[SubstFor[Sin[c*(a+b*x)],u,x],x],x],x,Sin[c*(a+b*x)]]] /;
FreeQ[{a,b,c},x] && FunctionOfQ[Sin[c*(a+b*x)],u,x,True]
```

- **Derivation: Integration by substitution**
- **Basis:** $f[\text{Sin}[z]]\, \text{Cot}[z] = \frac{f[\text{Sin}[z]]}{\text{Sin}[z]}\, \partial_z \text{Sin}[z]$
- **Rule:**

$$\int f[\text{Sin}[a+b\,x]]\, \text{Cot}[a+b\,x]\, dx \longrightarrow \frac{1}{b}\, \text{Subst}\left[\int \frac{f[x]}{x}\, dx,\, x,\, \text{Sin}[a+b\,x]\right]$$

- **Program code:**

```
Int[u_*Cot[c_.*(a_.+b_.*x_)],x_Symbol] :=
  Dist[1/(b*c),Subst[Int[Regularize[SubstFor[Sin[c*(a+b*x)],u,x]/x,x],x],x,Sin[c*(a+b*x)]]] /;
FreeQ[{a,b,c},x] && FunctionOfQ[Sin[c*(a+b*x)],u,x,True]
```

$$\int f[\text{Cos}[u]]\, \partial_x \text{Cos}[u]\, dx$$

- **Derivation: Integration by substitution**
- **Basis:** $f[\text{Cos}[z]]\,\text{Sin}[z] = -f[\text{Cos}[z]]\,\partial_z \text{Cos}[z]$
- **Rule:**

$$\int f[\text{Cos}[a+b\,x]]\,\text{Sin}[a+b\,x]\,dx \longrightarrow -\frac{1}{b}\,\text{Subst}\left[\int f[x]\,dx,\,x,\,\text{Cos}[a+b\,x]\right]$$

- **Program code:**

```
Int[u_*Sin[c_.*(a_.+b_.*x_)],x_Symbol] :=
  -Dist[1/(b*c),Subst[Int[Regularize[SubstFor[Cos[c*(a+b*x)],u,x],x],x],x,Cos[c*(a+b*x)]]] /;
FreeQ[{a,b,c},x] && FunctionOfQ[Cos[c*(a+b*x)],u,x,True]
```

- **Derivation: Integration by substitution**
- **Basis:** $f[\text{Cos}[z]]\,\text{Tan}[z] = -\frac{f[\text{Cos}[z]]}{\text{Cos}[z]}\,\partial_z \text{Cos}[z]$
- **Rule:**

$$\int f[\text{Cos}[a+b\,x]]\,\text{Tan}[a+b\,x]\,dx \longrightarrow -\frac{1}{b}\,\text{Subst}\left[\int \frac{f[x]}{x}\,dx,\,x,\,\text{Cos}[a+b\,x]\right]$$

- **Program code:**

```
Int[u_*Tan[c_.*(a_.+b_.*x_)],x_Symbol] :=
  -Dist[1/(b*c),Subst[Int[Regularize[SubstFor[Cos[c*(a+b*x)],u,x]/x,x],x],x,Cos[c*(a+b*x)]]] /;
FreeQ[{a,b,c},x] && FunctionOfQ[Cos[c*(a+b*x)],u,x,True]
```

$$\int f[\text{Cot}[u]]\, \partial_x \text{Cot}[u]\, dx$$

- **Derivation: Integration by substitution**
- **Basis:** $f[\text{Cot}[z]]\, \text{Csc}[z]^2 = -f[\text{Cot}[z]]\, \partial_z \text{Cot}[z]$
- **Rule:**

$$\int f[\text{Cot}[a+b\,x]]\, \text{Csc}[a+b\,x]^2\, dx \longrightarrow -\frac{1}{b}\, \text{Subst}\left[\int f[x]\, dx,\ x,\ \text{Cot}[a+b\,x]\right]$$

- **Program code:**

```
Int[u_*Csc[c_.*(a_.+b_.*x_)]^2,x_Symbol] :=
  -Dist[1/(b*c),Subst[Int[Regularize[SubstFor[Cot[c*(a+b*x)],u,x],x],x],x,Cot[c*(a+b*x)]]] /;
FreeQ[{a,b,c},x] && FunctionOfQ[Cot[c*(a+b*x)],u,x,True] && NonsumQ[u]
```

- **Derivation: Integration by substitution**
- **Basis:** If $n \in \mathbb{Z}$, then $f[\text{Cot}[z]]\, \text{Tan}[z]^n = -\frac{f[\text{Cot}[z]]}{\text{Cot}[z]^n\,(1+\text{Cot}[z]^2)}\, \partial_z \text{Cot}[z]$
- **Rule:** If $n \in \mathbb{Z}$, then

$$\int f[\text{Cot}[a+b\,x]]\, \text{Tan}[a+b\,x]^n\, dx \longrightarrow -\frac{1}{b}\, \text{Subst}\left[\int \frac{f[x]}{x^n\,(1+x^2)}\, dx,\ x,\ \text{Cot}[a+b\,x]\right]$$

- **Program code:**

```
Int[u_*Tan[c_.*(a_.+b_.*x_)]^n_.,x_Symbol] :=
  -Dist[1/(b*c),Subst[Int[Regularize[SubstFor[Cot[c*(a+b*x)],u,x]/(x^n*(1+x^2)),x],x],x,Cot[c*(a+b*x)
FreeQ[{a,b,c},x] && IntegerQ[n] && FunctionOfQ[Cot[c*(a+b*x)],u,x,True] && TryPureTanSubst[u*Tan[c*(a
```

- **Derivation: Integration by substitution**
- **Basis:** $f[\text{Cot}[z]] = -\frac{f[\text{Cot}[z]]}{1+\text{Cot}[z]^2}\,\partial_z \text{Cot}[z]$
- **Rule:**

$$\int f[\text{Cot}[a+b\,x]]\,dx \;\longrightarrow\; -\frac{1}{b}\,\text{Subst}\left[\int \frac{f[x]}{1+x^2}\,dx,\;x,\;\text{Cot}[a+b\,x]\right]$$

- **Program code:**

```
If[ShowSteps,

Int[u_,x_Symbol] :=
  Module[{v=FunctionOfTrig[u,x]},
  ShowStep["","Int[f[Cot[a+b*x]],x]","Subst[Int[f[x]/(1+x^2),x],x,Cot[a+b*x]]/b",Hold[
  Dist[-1/Coefficient[v,x,1],Subst[Int[Regularize[SubstFor[Cot[v],u,x]/(1+x^2),x],x],x,Cot[v]]]]] /;
 NotFalseQ[v] && FunctionOfQ[Cot[v],u,x,True] && TryPureTanSubst[u,x]] /;
SimplifyFlag,

Int[u_,x_Symbol] :=
  Module[{v=FunctionOfTrig[u,x]},
  Dist[-1/Coefficient[v,x,1],Subst[Int[Regularize[SubstFor[Cot[v],u,x]/(1+x^2),x],x],x,Cot[v]]] /;
 NotFalseQ[v] && FunctionOfQ[Cot[v],u,x,True] && TryPureTanSubst[u,x]]]
```

$$\int f[\mathrm{Tan}[u]]\, \partial_x \mathrm{Tan}[u]\, dx$$

- **Derivation: Integration by substitution**
- **Basis:** $f[\mathrm{Tan}[z]]\, \mathrm{Sec}[z]^2 = f[\mathrm{Tan}[z]]\, \partial_z \mathrm{Tan}[z]$
- **Rule:**

$$\int f[\mathrm{Tan}[a+b\,x]]\, \mathrm{Sec}[a+b\,x]^2\, dx \longrightarrow \frac{1}{b}\, \mathrm{Subst}\Big[\int f[x]\, dx,\ x,\ \mathrm{Tan}[a+b\,x]\Big]$$

- **Program code:**

```
Int[u_*Sec[c_.*(a_.+b_.*x_)]^2,x_Symbol] :=
  Dist[1/(b*c),Subst[Int[Regularize[SubstFor[Tan[c*(a+b*x)],u,x],x],x],x,Tan[c*(a+b*x)]]] /;
FreeQ[{a,b,c},x] && FunctionOfQ[Tan[c*(a+b*x)],u,x,True] && NonsumQ[u]
```

- **Derivation: Integration by substitution**
- **Basis:** If $n \in \mathbb{Z}$, then $f[\mathrm{Tan}[z]]\, \mathrm{Cot}[z]^n = \frac{f[\mathrm{Tan}[z]]}{\mathrm{Tan}[z]^n \left(1+\mathrm{Tan}[z]^2\right)}\, \partial_z \mathrm{Tan}[z]$
- **Rule:** If $n \in \mathbb{Z}$, then

$$\int f[\mathrm{Tan}[a+b\,x]]\, \mathrm{Cot}[a+b\,x]^n\, dx \longrightarrow \frac{1}{b}\, \mathrm{Subst}\Big[\int \frac{f[x]}{x^n \left(1+x^2\right)}\, dx,\ x,\ \mathrm{Tan}[a+b\,x]\Big]$$

- **Program code:**

```
Int[u_*Cot[c_.*(a_.+b_.*x_)]^n_.,x_Symbol] :=
  Dist[1/(b*c),Subst[Int[Regularize[SubstFor[Tan[c*(a+b*x)],u,x]/(x^n*(1+x^2)),x],x],x,Tan[c*(a+b*x)]
FreeQ[{a,b,c},x] && IntegerQ[n] && FunctionOfQ[Tan[c*(a+b*x)],u,x,True] && TryPureTanSubst[u*Cot[c*(a
```

- **Derivation: Integration by substitution**
- **Basis:** $f[\text{Tan}[z]] = \frac{f[\text{Tan}[z]]}{1+\text{Tan}[z]^2}\,\partial_z \text{Tan}[z]$
- **Rule:**

$$\int f[\text{Tan}[a+b\,x]]\,dx \longrightarrow \frac{1}{b}\,\text{Subst}\left[\int \frac{f[x]}{1+x^2}\,dx,\,x,\,\text{Tan}[a+b\,x]\right]$$

- **Program code:**

```
If[ShowSteps,

Int[u_,x_Symbol] :=
  Module[{v=FunctionOfTrig[u,x]},
  ShowStep["","Int[f[Tan[a+b*x]],x]","Subst[Int[f[x]/(1+x^2),x],x,Tan[a+b*x]]/b",Hold[
  Dist[1/Coefficient[v,x,1],Subst[Int[Regularize[SubstFor[Tan[v],u,x]/(1+x^2),x],x],x,Tan[v]]]]] /;
 NotFalseQ[v] && FunctionOfQ[Tan[v],u,x,True] && TryPureTanSubst[u,x]] /;
SimplifyFlag,

Int[u_,x_Symbol] :=
  Module[{v=FunctionOfTrig[u,x]},
  Dist[1/Coefficient[v,x,1],Subst[Int[Regularize[SubstFor[Tan[v],u,x]/(1+x^2),x],x],x,Tan[v]]] /;
 NotFalseQ[v] && FunctionOfQ[Tan[v],u,x,True] && TryPureTanSubst[u,x]]]
```

$$\int u\,(a+b\,\mathrm{Sin}[c+d\,x])^n\,dx$$

- **Note: This rule should be moved just before the tangent $\theta/2$ substitution rules for linear trigonometric expressions**

- **Derivation: Piecewise constant extraction and algebraic expansion**

- **Basis: If** $a^2-b^2=0$, **then** $\partial_z \frac{\sqrt{a+b\,\mathrm{Sin}[z]}}{\mathrm{Cos}\left[\frac{z}{2}\right]+\frac{a}{b}\,\mathrm{Sin}\left[\frac{z}{2}\right]}=0$

- **Rule: If** $a^2-b^2=0\ \bigwedge\ n-\frac{1}{2}\in\mathbb{Z}$, **then**

$$\int u\,(a+b\,\mathrm{Sin}[c+d\,x])^n\,dx\ \longrightarrow$$

$$\frac{\sqrt{a+b\,\mathrm{Sin}[c+d\,x]}}{\mathrm{Cos}\left[\frac{c+d\,x}{2}\right]+\frac{a}{b}\,\mathrm{Sin}\left[\frac{c+d\,x}{2}\right]}\left(\int u\,\mathrm{Cos}\left[\frac{c+d\,x}{2}\right]\,(a+b\,\mathrm{Sin}[c+d\,x])^{n-\frac{1}{2}}\,dx+\frac{a}{b}\int u\,\mathrm{Sin}\left[\frac{c+d\,x}{2}\right]\,(a+b\,\mathrm{Sin}[c+d\,x])^{n-\frac{1}{2}}\,dx\right)$$

- **Program code:**

```
(* Int[u_*(a_+b_.*Sin[c_.+d_.*x_])^n_,x_Symbol] :=
  Sqrt[a+b*Sin[c+d*x]]/(Cos[c/2+d*x/2]+a/b*Sin[c/2+d*x/2])*
    (Int[u*Cos[c/2+d*x/2]*(a+b*Sin[c+d*x])^(n-1/2),x] +
     a/b*Int[u*Sin[c/2+d*x/2]*(a+b*Sin[c+d*x])^(n-1/2),x]) /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2-b^2] && IntegerQ[n-1/2] *)
```

$$\int u\,(a+b\,\mathrm{Cos}[c+d\,x])^n\,dx$$

- **Note: These rules should be moved just before the tangent $\theta/2$ substitution rules for linear trigonometric expressions**

- **Derivation: Piecewise constant extraction**

- **Basis:** $\partial_z \frac{\sqrt{a+a\,\mathrm{Cos}[z]}}{\mathrm{Cos}\left[\frac{z}{2}\right]} = 0$

- **Rule: If** $n-\frac{1}{2}\in\mathbb{Z}$**, then**

$$\int u\,(a+a\,\mathrm{Cos}[c+d\,x])^n\,dx \longrightarrow \frac{\sqrt{a+a\,\mathrm{Cos}[c+d\,x]}}{\mathrm{Cos}\left[\frac{c+d\,x}{2}\right]}\int u\,\mathrm{Cos}\left[\frac{c+d\,x}{2}\right]\,(a+a\,\mathrm{Cos}[c+d\,x])^{n-\frac{1}{2}}\,dx$$

- **Program code:**

```
(* Int[u_*(a_+b_.*Cos[c_.+d_.*x_])^n_,x_Symbol] :=
  Sqrt[a+b*Cos[c+d*x]]/Cos[c/2+d*x/2]*Int[u*Cos[c/2+d*x/2]*(a+b*Cos[c+d*x])^(n-1/2),x] /;
FreeQ[{a,b,c,d},x] && ZeroQ[a-b] && IntegerQ[n-1/2] *)
```

- **Derivation: Piecewise constant extraction**

- **Basis:** $\partial_z \frac{\sqrt{a-a\,\mathrm{Cos}[z]}}{\mathrm{Sin}\left[\frac{z}{2}\right]} = 0$

- **Rule: If** $n-\frac{1}{2}\in\mathbb{Z}$**, then**

$$\int u\,(a-a\,\mathrm{Cos}[c+d\,x])^n\,dx \longrightarrow \frac{\sqrt{a-a\,\mathrm{Cos}[c+d\,x]}}{\mathrm{Sin}\left[\frac{c+d\,x}{2}\right]}\int u\,\mathrm{Sin}\left[\frac{c+d\,x}{2}\right]\,(a-a\,\mathrm{Cos}[c+d\,x])^{n-\frac{1}{2}}\,dx$$

- **Program code:**

```
(* Int[u_*(a_+b_.*Cos[c_.+d_.*x_])^n_,x_Symbol] :=
  Sqrt[a+b*Cos[c+d*x]]/Sin[c/2+d*x/2]*Int[u*Sin[c/2+d*x/2]*(a+b*Cos[c+d*x])^(n-1/2),x] /;
FreeQ[{a,b,c,d},x] && ZeroQ[a+b] && IntegerQ[n-1/2] *)
```

$$\int u\,(a+b\,\mathrm{Cos}[d+e\,x]+c\,\mathrm{Sin}[d+e\,x])^n\,dx$$

- **Note: This rule should be moved just before the tangent $\theta/2$ substitution rules for linear trigonometric expressions**
- **Derivation: Piecewise constant extraction and algebraic expansion**
- **Basis: If** $a^2-b^2-c^2=0$, **then** $\partial_z \frac{\sqrt{a+b\,\mathrm{Cos}[z]+c\,\mathrm{Sin}[z]}}{c\,\mathrm{Cos}\left[\frac{z}{2}\right]+(a-b)\,\mathrm{Sin}\left[\frac{z}{2}\right]}=0$
- **Rule: If** $a^2-b^2-c^2=0\bigwedge n-\frac{1}{2}\in\mathbb{Z}$, **then**

$$\int u\,(a+b\,\mathrm{Cos}[d+e\,x]+c\,\mathrm{Sin}[d+e\,x])^n\,dx\;\longrightarrow$$

$$\frac{c\sqrt{a+b\,\mathrm{Cos}[d+e\,x]+c\,\mathrm{Sin}[d+e\,x]}}{c\,\mathrm{Cos}\left[\frac{d+e\,x}{2}\right]+(a-b)\,\mathrm{Sin}\left[\frac{d+e\,x}{2}\right]}\int u\,\mathrm{Cos}\left[\frac{d+e\,x}{2}\right](a+b\,\mathrm{Cos}[d+e\,x]+c\,\mathrm{Sin}[d+e\,x])^{n-\frac{1}{2}}\,dx+$$
$$\frac{(a-b)\sqrt{a+b\,\mathrm{Cos}[d+e\,x]+c\,\mathrm{Sin}[d+e\,x]}}{c\,\mathrm{Cos}\left[\frac{d+e\,x}{2}\right]+(a-b)\,\mathrm{Sin}\left[\frac{d+e\,x}{2}\right]}\int u\,\mathrm{Sin}\left[\frac{d+e\,x}{2}\right](a+b\,\mathrm{Cos}[d+e\,x]+c\,\mathrm{Sin}[d+e\,x])^{n-\frac{1}{2}}\,dx$$

- **Program code:**

```
(* Int[u_*(a_+b_.*Cos[d_.+e_.*x_]+c_.*Sin[d_.+e_.*x_])^n_,x_Symbol] :=
  Sqrt[a+b*Cos[d+e*x]+c*Sin[d+e*x]]/(c*Cos[(d+e*x)/2]+(a-b)*Sin[(d+e*x)/2])*
    Dist[c,Int[u*Cos[(d+e*x)/2]*(a+b*Cos[d+e*x]+c*Sin[d+e*x])^(n-1/2),x]] +
  Sqrt[a+b*Cos[d+e*x]+c*Sin[d+e*x]]/(c*Cos[(d+e*x)/2]+(a-b)*Sin[(d+e*x)/2])*
    Dist[a-b,Int[u*Sin[(d+e*x)/2]*(a+b*Cos[d+e*x]+c*Sin[d+e*x])^(n-1/2),x]] /;
FreeQ[{a,b,c,d,e},x] && ZeroQ[a^2-b^2-c^2] && IntegerQ[n-1/2] *)
```