Integration Rules for

$$\int \left(\sin^j(z)\right)^m \left(a+b\sin^k(z)\right)^n \, dz \text{ when } j^2=1 \bigwedge k^2=1$$

Domain Map



Legend:

- The rule number in a colored region indicates the rule to use for integrals in that region.
- The rule number next to a colored line indicates the rule to use for integrals on that line.
- A white region or line indicates there is no rule for integrals in that region or on that line.
- A solid black line indicates integrals on that line are handled by rules in another section.
- A dashed black line on the border of a region indicates integrals on that border are handled by the rule for that region.
- The arrow(s) following a rule number indicates the direction the rule drives integrands in the n×m exponent plane.
- A ⋄ following a rule number indicates the rule transforms the integrand into a form handled by another section.
- A red (stop) disk indicates the terminal rule to use for the point at the center of the disk.
- A cyan disk indicates the non-terminal rule to use for the point at the center of the disk.

Integration Rules for

$\int \left(\sin^j(z)\right)^m \, dz$ when $j^2 = 1$

Rule a: $\int \text{Sin}[c + d\,x]^j \, dx$

- **Reference: G&R 2.01.5, CRC 290, A&S 4.3.113**
- **Derivation: Rule 8b with m = 1 and j = 1**
- **Note: This rule is an unnecessary special case of rule 8b, but it saves a step.**
- **Rule a1:**

$$\int \text{Sin}[c + d\,x] \, dx \longrightarrow -\frac{\text{Cos}[c + d\,x]}{d}$$

- **Program code:**

```
Int[sin[c_.+d_.*x_],x_Symbol] :=
  -Cos[c+d*x]/d /;
FreeQ[{c,d},x]
```

- **Reference: G&R 2.01.6, CRC 291, A&S 4.3.114**

```
Int[Cos[a_.+b_.*x_],x_Symbol] :=
  Sin[a+b*x]/b /;
FreeQ[{a,b},x]
```

- **Reference: G&R 2.526.1, CRC 295, A&S 4.3.116'**
- **Derivation: Integration by substitution**
- **Basis:** $\text{Csc}[z] = -\frac{1}{1-\text{Cos}[z]^2} \text{Cos}'[z]$
- **Rule a2:**

$$\int \text{Csc}[c + d\,x] \, dx \longrightarrow -\frac{\text{ArcTanh}[\text{Cos}[c + d\,x]]}{d}$$

- **Program code:**

```
Int[1/sin[c_.+d_.*x_],x_Symbol] :=
  -ArcTanh[Cos[c+d*x]]/d /;
FreeQ[{c,d},x]
```

- **Reference: G&R 2.526.9', CRC 294', A&S 4.3.117'**

```
Int[Sec[a_.+b_.*x_],x_Symbol] :=
  ArcTanh[Sin[a+b*x]]/b /;
FreeQ[{a,b},x]
```

Rule b: $\int \mathrm{Sin}[c+d\,x]^{2\,j}\,dx$

- **Reference: G&R 2.513.5, CRC 296**
- **Derivation: Rule 8b with $m = 2$ and $j = 1$**
- **Note: This rule is an unnecessary special case of rule 8b, but it saves a step.**
- **Rule b1:**

$$\int \mathrm{Sin}[c+d\,x]^2\,dx \longrightarrow \frac{x}{2}-\frac{\mathrm{Cos}[c+d\,x]\,\mathrm{Sin}[c+d\,x]}{2\,d}$$

- **Program code:**

```
Int[sin[c_.+d_.*x_]^2,x_Symbol] :=
  x/2 - Cos[c+d*x]*Sin[c+d*x]/(2*d) /;
FreeQ[{c,d},x]
```

- **Reference: G&R 2.513.11, CRC 302**

```
Int[Cos[a_.+b_.*x_]^2,x_Symbol] :=
  x/2 + Cos[a+b*x]*Sin[a+b*x]/(2*b) /;
FreeQ[{a,b},x]
```

- **Reference: G&R 2.526.2, CRC 308**
- **Derivation: Rule 7b with $m = -2$ and $j = 1$**
- **Note: This rule is an unnecessary special case of rule 7b, but it saves a step.**
- **Rule b2:**

$$\int \mathrm{Csc}[c+d\,x]^2\,dx \longrightarrow -\frac{\mathrm{Cot}[c+d\,x]}{d}$$

- **Program code:**

```
Int[1/sin[c_.+d_.*x_]^2,x_Symbol] :=
  -Cot[c+d*x]/d /;
FreeQ[{c,d},x]
```

- **Reference: G&R 2.526.10, CRC 312**

```
Int[Sec[a_.+b_.*x_]^2,x_Symbol] :=
  Tan[a+b*x]/b /;
FreeQ[{a,b},x]
```

Rules 7 – 8: $\int \left(\operatorname{Sin}[c+d\,x]^j\right)^m \,dx$

- **Derivation: Integration by substitution**
- **Basis: If** $\frac{m}{2} \in \mathbb{Z}$, **then** $\operatorname{Csc}[z]^m = -\left(1+\operatorname{Cot}[z]^2\right)^{\frac{m-2}{2}} \operatorname{Cot}'[z]$
- **Note: This rule is used for odd** $m < -2$ **since it requires fewer steps and results in simpler antiderivatives than rule 7b.**
- **Rule 7a: If** $\frac{m}{2} \in \mathbb{Z} \bigwedge m < -2$, **then**

$$\int \operatorname{Sin}[c+d\,x]^m \,dx \longrightarrow -\frac{1}{d} \operatorname{Subst}\left[\int \left(1+x^2\right)^{\frac{-m-2}{2}} dx,\ x,\ \operatorname{Cot}[c+d\,x]\right]$$

- **Program code:**

```
Int[sin[c_.+d_.*x_]^m_,x_Symbol] :=
  Dist[-1/d,Subst[Int[Expand[(1+x^2)^((-m-2)/2),x],x],x,Cot[c+d*x]]] /;
FreeQ[{c,d},x] && EvenQ[m] && m<-2
```

```
Int[Sec[a_.+b_.*x_]^n_,x_Symbol] :=
  Dist[1/b,Subst[Int[Regularize[(1+x^2)^((n-2)/2),x],x],x,Tan[a+b*x]]] /;
FreeQ[{a,b},x] && EvenQ[n] && n>2
```

- **Reference: G&R 2.510.3, CRC 309**
- **Reference: G&R 2.552.3 special case when** $a = 0$
- **Derivation: Rule 5b with** $a = 1$, $b = 0$, $k = j$ **and** $n = 0$
- **Rule 7b: If** $j^2 = 1 \bigwedge \frac{m}{2} \notin \mathbb{Z} \bigwedge m < -1$, **then**

$$\int \left(\operatorname{Sin}[c+d\,x]^j\right)^m dx \longrightarrow \frac{2 \operatorname{Cos}[c+d\,x] \left(\operatorname{Sin}[c+d\,x]^j\right)^{m+1}}{d\,(2\,m+j+1)} + \frac{2\,m+j+3}{2\,m+j+1} \int \left(\operatorname{Sin}[c+d\,x]^j\right)^{m+2} dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_,x_Symbol] :=
  2*Cos[c+d*x]*(Sin[c+d*x]^j)^(m+1)/(d*(2*m+j+1)) +
   Dist[(2*m+j+3)/(2*m+j+1),Int[(sin[c+d*x]^j)^(m+2),x]] /;
FreeQ[{c,d},x] && OneQ[j^2] && Not[EvenQ[m]] && RationalQ[m] && m<-1
```

- **Reference: G&R 2.510.6, CRC 313**

```
Int[Cos[a_.+b_.*x_]^n_,x_Symbol] :=
  -Sin[a+b*x]*Cos[a+b*x]^(n+1)/(b*(n+1)) +
  Dist[(n+2)/(n+1),Int[Cos[a+b*x]^(n+2),x]] /;
FreeQ[{a,b},x] && RationalQ[n] && n<-1
```

```
Int[Sec[a_.+b_.*x_]^n_,x_Symbol] :=
  Sin[a+b*x]*Sec[a+b*x]^(n-1)/(b*(n-1)) +
  Dist[(n-2)/(n-1),Int[Sec[a+b*x]^(n-2),x]] /;
FreeQ[{a,b},x] && RationalQ[n] && n>1
```

- **Derivation: Integration by substitution**
- **Basis: If** $\frac{m-1}{2} \in \mathbb{Z}$**, then** $\text{Sin}[z]^m = -\left(1 - \text{Cos}[z]^2\right)^{\frac{m-1}{2}} \text{Cos}'[z]$
- **Note: This rule is used for odd** $m > 1$ **since it requires fewer steps and results in simpler antiderivatives than rule 8b.**
- **Rule 8a: If** $\frac{m-1}{2} \in \mathbb{Z} \bigwedge m > 1$**, then**

$$\int \text{Sin}[c + d\,x]^m \, dx \longrightarrow -\frac{1}{d} \text{Subst}\left[\int \left(1 - x^2\right)^{\frac{m-1}{2}} dx,\ x,\ \text{Cos}[c + d\,x]\right]$$

- **Program code:**

```
Int[sin[c_.+d_.*x_]^m_,x_Symbol] :=
  Dist[-1/d,Subst[Int[Expand[(1-x^2)^((m-1)/2),x],x],x,Cos[c+d*x]]] /;
FreeQ[{c,d},x] && OddQ[m] && m>1
```

```
Int[Cos[a_.+b_.*x_]^n_,x_Symbol] :=
  Dist[1/b,Subst[Int[Expand[(1-x^2)^((n-1)/2),x],x],x,Sin[a+b*x]]] /;
FreeQ[{a,b},x] && OddQ[n] && n>1
```

- **Reference: G&R 2.510.2, CRC 299**
- **Reference: G&R 2.552.3 inverted special case when** $a = 0$
- **Derivation: Rule 2b with** $k = j$ **and** $n = 0$
- **Rule 8b: If** $j^2 = 1 \bigwedge \frac{m-1}{2} \notin \mathbb{Z} \bigwedge m > 1$**, then**

$$\int \left(\text{Sin}[c + d\,x]^j\right)^m dx \longrightarrow -\frac{2\,\text{Cos}[c + d\,x]\left(\text{Sin}[c + d\,x]^j\right)^{m-1}}{d\,(2\,m + j - 1)} + \frac{2\,m + j - 3}{2\,m + j - 1} \int \left(\text{Sin}[c + d\,x]^j\right)^{m-2} dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_,x_Symbol] :=
  -2*Cos[c+d*x]*(Sin[c+d*x]^j)^(m-1)/(d*(2*m+j-1)) +
   Dist[(2*m+j-3)/(2*m+j-1),Int[(sin[c+d*x]^j)^(m-2),x]] /;
FreeQ[{c,d},x] && OneQ[j^2] && Not[OddQ[m]] && RationalQ[m] && m>1
```

- **Reference: G&R 2.510.5, CRC 305**

```
Int[Cos[a_.+b_.*x_]^n_,x_Symbol] :=
  Sin[a+b*x]*Cos[a+b*x]^(n-1)/(b*n) +
  Dist[(n-1)/n,Int[Cos[a+b*x]^(n-2),x]] /;
FreeQ[{a,b},x] && RationalQ[n] && n>1
```

```
Int[Sec[a_.+b_.*x_]^n_,x_Symbol] :=
  -Sin[a+b*x]*Sec[a+b*x]^(n+1)/(b*n) +
  Dist[(n+1)/n,Int[Sec[a+b*x]^(n+2),x]] /;
FreeQ[{a,b},x] && RationalQ[n] && n<-1
```

Integration Rules for

$\int (a + b \sin^k(z))^n \, dz$ when $k^2 = 1$

Rule c: $\int \frac{1}{a + b \operatorname{Sin}[c + d x]^k} \, dx$

- **Note: Although this rule produces a slightly more complicated antiderivative than rule c2 and c4, it is continuous provided $a^2 - b^2 > 0$.**

- **Rule c1: If $a^2 - b^2 > 0$, then**

$$\int \frac{1}{a + b \operatorname{Sin}[c + d x]} \, dx \longrightarrow \frac{x}{\sqrt{a^2 - b^2}} + \frac{2}{d \sqrt{a^2 - b^2}} \operatorname{ArcTan}\left[\frac{b \operatorname{Cos}[c + d x]}{a + \sqrt{a^2 - b^2} + b \operatorname{Sin}[c + d x]}\right]$$

- **Program code:**

```
Int[1/(a_+b_.*sin[c_.+d_.*x_]),x_Symbol] :=
  x/Rt[a^2-b^2,2] +
  2/(d*Rt[a^2-b^2,2])*ArcTan[Sim[b*Cos[c+d*x]/(a+Rt[a^2-b^2,2]+b*Sin[c + d*x])]] /;
FreeQ[{a,b,c,d},x] && PositiveQ[a^2-b^2]
```

```
Int[1/(a_+b_.*Cos[c_.+d_.*x_]),x_Symbol] :=
  x/Rt[a^2-b^2,2] - 2/(d*Rt[a^2-b^2,2])*ArcTan[Sim[b*Sin[c+d*x]/(a+Rt[a^2-b^2,2]+b*Cos[c + d*x])]] /
FreeQ[{a,b,c,d},x] && PositiveQ[a^2-b^2]
```

- **Reference: G&R 2.553.3a, A&S 4.3.133a**

- **Note: Although nonessential, this rule produces a slightly simpler antiderivative than rule c3.**

- **Rule c2: If $a^2 - b^2 > 0$, then**

$$\int \frac{1}{a + b \operatorname{Cos}[c + d x]} \, dx \longrightarrow \frac{2}{d \sqrt{a^2 - b^2}} \operatorname{ArcTan}\left[\frac{(a - b) \operatorname{Tan}\left[\frac{1}{2} (c + d x)\right]}{\sqrt{a^2 - b^2}}\right]$$

- **Program code:**

```
Int[1/(a_+b_.*sin[c_.+Pi/2+d_.*x_]),x_Symbol] :=
  2*ArcTan[(a-b)*Tan[(c+d*x)/2]/Rt[a^2-b^2,2]]/(d*Rt[a^2-b^2,2]) /;
FreeQ[{a,b,c,d},x] && PosQ[a^2-b^2]
```

- **Reference: G&R 2.551.3a, A&S 4.3.131a**
- **Rule c3: If $a^2 - b^2 > 0$, then**

$$\int \frac{1}{a + b \, \mathrm{Sin}[c + d\, x]} \, dx \;\longrightarrow\; \frac{2}{d \sqrt{a^2 - b^2}} \, \mathrm{ArcTan}\left[\frac{b + a \, \mathrm{Tan}\left[\frac{1}{2}\,(c + d\, x)\right]}{\sqrt{a^2 - b^2}}\right]$$

- **Program code:**

```
Int[1/(a_+b_.*sin[c_.+d_.*x_]),x_Symbol] :=
  2*ArcTan[(b+a*Tan[(c+d*x)/2])/Rt[a^2-b^2,2]]/(d*Rt[a^2-b^2,2]) /;
FreeQ[{a,b,c,d},x] && PosQ[a^2-b^2]
```

- **Reference: G&R 2.553.3b', A&S 4.3.133b'**
- **Note: Although nonessential, this rule produces a slightly simpler antiderivative than rule c5.**
- **Rule c4: If $\neg\,(a^2 - b^2 > 0)$, then**

$$\int \frac{1}{a + b \, \mathrm{Cos}[c + d\, x]} \, dx \;\longrightarrow\; -\frac{2}{d \sqrt{b^2 - a^2}} \, \mathrm{ArcTanh}\left[\frac{(a - b)\, \mathrm{Tan}\left[\frac{1}{2}\,(c + d\, x)\right]}{\sqrt{b^2 - a^2}}\right]$$

- **Program code:**

```
Int[1/(a_+b_.*sin[c_.+Pi/2+d_.*x_]),x_Symbol] :=
  -2*ArcTanh[(a-b)*Tan[(c+d*x)/2]/Rt[b^2-a^2,2]]/(d*Rt[b^2-a^2,2]) /;
FreeQ[{a,b,c,d},x] && NegQ[a^2-b^2]
```

- **Reference: G&R 2.551.3b', A&S 4.3.131b'**
- **Rule c5: If $\neg\,(a^2 - b^2 > 0)$, then**

$$\int \frac{1}{a + b \, \mathrm{Sin}[c + d\, x]} \, dx \;\longrightarrow\; -\frac{2}{d \sqrt{b^2 - a^2}} \, \mathrm{ArcTanh}\left[\frac{b + a \, \mathrm{Tan}\left[\frac{1}{2}\,(c + d\, x)\right]}{\sqrt{b^2 - a^2}}\right]$$

- **Program code:**

```
Int[1/(a_+b_.*sin[c_.+d_.*x_]),x_Symbol] :=
  -2*ArcTanh[(b+a*Tan[(c+d*x)/2])/Rt[b^2-a^2,2]]/(d*Rt[b^2-a^2,2]) /;
FreeQ[{a,b,c,d},x] && NegQ[a^2-b^2]
```

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{1}{a+b/z} = \frac{1}{a} - \frac{b}{a\,(b+a\,z)}$
- **Rule c6: If** $a^2 - b^2 \neq 0$**, then**

$$\int \frac{1}{a + b\,\mathrm{Csc}[c + d\,x]}\,dx \longrightarrow \frac{x}{a} - \frac{b}{a}\int \frac{1}{b + a\,\mathrm{Sin}[c + d\,x]}\,dx$$

- **Program code:**

```
Int[1/(a_+b_.*sin[c_.+d_.*x_]^(-1)),x_Symbol] :=
  x/a - Dist[b/a,Int[1/(b+a*sin[c+d*x]),x]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2-b^2]
```

Rule d: $\int \left(a + b \operatorname{Sin}[c + d x]^k\right)^2 \, dx$

- **Derivation: Rule 12 with $m = 0$**
- **Rule d: If $k^2 = 1$, then**

$$\int \left(a + b \operatorname{Sin}[c + d x]^k\right)^2 \, dx \longrightarrow \left(a^2 + \frac{k+1}{k+3} b^2\right) x - \frac{2 b^2 \operatorname{Cos}[c + d x] \operatorname{Sin}[c + d x]^k}{d (k + 3)} + 2 a b \int \operatorname{Sin}[c + d x]^k \, dx$$

- **Program code:**

```
Int[(a_+b_.*sin[c_.+d_.*x_]^k_.)^2,x_Symbol] :=
  (a^2+(k+1)/(k+3)*b^2)*x - 2*b^2*Cos[c+d*x]*Sin[c+d*x]^k/(d*(k+3)) + 2*a*b*Int[sin[c+d*x]^k,x] /;
FreeQ[{a,b,c,d},x] && OneQ[k^2]
```

Rule e: $\int \sqrt{a + b \,\text{Sin}[c + d\, x]}\, dx$

- **Basis:** $\partial_z \text{EllipticE}[z, n] = \sqrt{1 - n\,\text{Sin}[z]^2}$
- **Basis:** $1 - \frac{2\,b}{a+b}\,\text{Sin}\left[\frac{c+d\,x}{2} - \frac{\pi}{4}\right]^2 = \frac{a}{a+b} + \frac{b\,\text{Sin}[c+d\,x]}{a+b}$
- **Rule e1: If** $a^2 - b^2 \neq 0 \,\land\, a + b > 0$**, then**

$$\int \sqrt{a + b\,\text{Sin}[c + d\,x]}\, dx \longrightarrow -\frac{2\sqrt{a+b}}{d}\,\text{EllipticE}\left[\frac{\pi}{4} - \frac{c + d\,x}{2},\, \frac{2\,b}{a + b}\right]$$

- **Program code:**

```
Int[Sqrt[a_.+b_.*sin[c_.+Pi/2+d_.*x_]],x_Symbol] :=
  2*Sqrt[Sim[a+b]]/d*EllipticE[(c+d*x)/2,Sim[2*b/(a+b)]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2-b^2] && PositiveQ[a+b]
```

```
Int[Sqrt[a_.+b_.*sin[c_.+d_.*x_]],x_Symbol] :=
  -2*Sqrt[Sim[a+b]]/d*EllipticE[Pi/4-(c+d*x)/2,Sim[2*b/(a+b)]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2-b^2] && PositiveQ[a+b]
```

- **Derivation: Piecewise constant extraction**
- **Basis:** $\partial_z \frac{\sqrt{\frac{a}{a+b} + \frac{b}{a+b}\,f[z]}}{\sqrt{a+b\,f[z]}} = 0$
- **Note: Since** $\frac{a}{a+b} + \frac{b}{a+b} = 1 > 0$**, rule e1 applies to the resulting integrand.**
- **Rule e2: If** $a^2 - b^2 \neq 0 \,\land\, \neg\,(a + b > 0)$**, then**

$$\int \sqrt{a + b\,\text{Sin}[c + d\,x]}\, dx \longrightarrow \frac{(a + b)\sqrt{\frac{a}{a+b} + \frac{b}{a+b}\,\text{Sin}[c + d\,x]}}{\sqrt{a + b\,\text{Sin}[c + d\,x]}} \int \sqrt{\frac{a}{a + b} + \frac{b}{a + b}\,\text{Sin}[c + d\,x]}\, dx$$

- **Program code:**

```
Int[Sqrt[a_.+b_.*sin[c_.+d_.*x_]],x_Symbol] :=
  (a+b)*Sqrt[(a+b*Sin[c+d*x])/(a+b)]/Sqrt[a+b*Sin[c+d*x]]*Int[Sqrt[a/(a+b)+b/(a+b)*sin[c+d*x]],x] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2-b^2] && Not[PositiveQ[a+b]]
```

Rule f: $\int \frac{1}{\sqrt{a + b \operatorname{Sin}[c + d x]}} \, dx$

- **Basis:** $\partial_x \operatorname{EllipticF}[x, n] = \frac{1}{\sqrt{1 - n \operatorname{Sin}[x]^2}}$
- **Basis:** $1 - \frac{2 b}{a+b} \operatorname{Sin}\left[\frac{c+d x}{2} - \frac{\pi}{4}\right]^2 = \frac{a}{a+b} + \frac{b \operatorname{Sin}[c+d x]}{a+b}$
- **Rule f1: If** $a^2 - b^2 \neq 0 \wedge a + b > 0$**, then**

$$\int \frac{1}{\sqrt{a + b \operatorname{Sin}[c + d x]}} \, dx \longrightarrow -\frac{2}{d \sqrt{a + b}} \operatorname{EllipticF}\left[\frac{\pi}{4} - \frac{c + d x}{2}, \frac{2 b}{a + b}\right]$$

- **Program code:**

```
Int[1/Sqrt[a_.+b_.*sin[c_.+Pi/2+d_.*x_]],x_Symbol] :=
  2/(d*Sqrt[Sim[a+b]])*EllipticF[(c+d*x)/2,Sim[2*b/(a+b)]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2-b^2] && PositiveQ[a+b]
```

```
Int[1/Sqrt[a_.+b_.*sin[c_.+d_.*x_]],x_Symbol] :=
  -2/(d*Sqrt[Sim[a+b]])*EllipticF[Pi/4-(c+d*x)/2,Sim[2*b/(a+b)]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2-b^2] && PositiveQ[a+b]
```

- **Derivation: Piecewise constant extraction**
- **Basis:** $\partial_z \frac{\sqrt{\frac{a+b f[z]}{a+b}}}{\sqrt{a+b f[z]}} = 0$
- **Note: Since** $\frac{a}{a+b} + \frac{b}{a+b} = 1 > 0$**, rule f1 applies to the resulting integrand.**
- **Rule f2: If** $a^2 - b^2 \neq 0 \wedge \neg (a + b > 0)$**, then**

$$\int \frac{1}{\sqrt{a + b \operatorname{Sin}[c + d x]}} \, dx \longrightarrow \frac{\sqrt{\frac{a}{a+b} + \frac{b}{a+b} \operatorname{Sin}[c + d x]}}{\sqrt{a + b \operatorname{Sin}[c + d x]}} \int \frac{1}{\sqrt{\frac{a}{a+b} + \frac{b}{a+b} \operatorname{Sin}[c + d x]}} \, dx$$

- **Program code:**

```
Int[1/Sqrt[a_.+b_.*sin[c_.+d_.*x_]],x_Symbol] :=
  Sqrt[(a+b*Sin[c+d*x])/(a+b)]/Sqrt[a+b*Sin[c+d*x]]*Int[1/Sqrt[a/(a+b)+b/(a+b)*sin[c+d*x]],x] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2-b^2] && Not[PositiveQ[a+b]]
```

$$\int (a + b\,\mathrm{Csc}[c + d\,x])^{n/2}\,dx$$

- **Derivation: Piecewise constant extraction**
- **Basis:** $\partial_z \left(f[z]^{n/2}\,(b / f[z])^{n/2}\right) = 0$
- **Rule: If** $n^2 = 1$, **then**

$$\int (b\,\mathrm{Csc}[c + d\,x])^{n/2}\,dx \longrightarrow (\mathrm{Sin}[c + d\,x])^{n/2}\,(b\,\mathrm{Csc}[c + d\,x])^{n/2} \int \frac{1}{\mathrm{Sin}[c + d\,x]^{n/2}}\,dx$$

- **Program code:**

```
Int[(b_.*sin[c_.+d_.*x_]^(-1))^n_,x_Symbol] :=
  Dist[Sin[c+d*x]^n*(b*Csc[c+d*x])^n,Int[1/sin[c+d*x]^n,x]] /;
FreeQ[{b,c,d},x] && ZeroQ[n^2-1/4]
```

- **Derivation: Piecewise constant extraction**
- **Basis:** $\partial_z \frac{\sqrt{b+a\,f[z]}}{\sqrt{f[z]}\,\sqrt{a+b/f[z]}} = 0$
- **Rule: If** $a^2 - b^2 \neq 0 \bigwedge n - \frac{1}{2} \in \mathbb{Z} \bigwedge -2 < n < 2$, **then**

$$\int (a + b\,\mathrm{Csc}[c + d\,x])^n\,dx \longrightarrow \frac{\sqrt{b + a\,\mathrm{Sin}[c + d\,x]}}{\sqrt{\mathrm{Sin}[c + d\,x]}\,\sqrt{a + b\,\mathrm{Csc}[c + d\,x]}} \int \frac{(b + a\,\mathrm{Sin}[c + d\,x])^n}{\mathrm{Sin}[c + d\,x]^n}\,dx$$

- **Program code:**

```
Int[(a_+b_.*sin[c_.+d_.*x_]^(-1))^n_,x_Symbol] :=
  Dist[Sqrt[b+a*Sin[c+d*x]]/(Sqrt[Sin[c+d*x]]*Sqrt[a+b*Csc[c+d*x]]),
    Int[(b+a*sin[c+d*x])^n/sin[c+d*x]^n,x]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2-b^2] && IntegerQ[n-1/2] && -2<n<2
```

Rules 17 – 18: $\int (a + b \,\mathrm{Csc}[c + d\, x])^n \,dx$

- **Derivation: Rule 6 with** $m = 0$ **and** $k = -1$
- **Rule 17: If** $a^2 - b^2 \neq 0 \land n < -1$**, then**

$$\int (a + b \,\mathrm{Csc}[c + d\, x])^n \,dx \longrightarrow$$

$$\frac{b^2 \,\mathrm{Cot}[c + d\, x] \,(a + b \,\mathrm{Csc}[c + d\, x])^{n+1}}{a\, d\, (n + 1) \left(a^2 - b^2\right)} + \frac{1}{a\, (n + 1) \left(a^2 - b^2\right)} \cdot$$

$$\int \left(\left(a^2 - b^2\right) (n + 1) - a\, b\, (n + 1) \,\mathrm{Csc}[c + d\, x] + b^2\, (n + 2) \,\mathrm{Csc}[c + d\, x]^2\right) (a + b \,\mathrm{Csc}[c + d\, x])^{n+1} \,dx$$

- **Program code:**

```
Int[(a_+b_.*sin[c_.+d_.*x_]^(-1))^n_,x_Symbol] :=
  b^2*Cot[c+d*x]*(a+b*Csc[c+d*x])^(n+1)/(a*d*(n+1)*(a^2-b^2)) +
  Dist[1/(a*(n+1)*(a^2-b^2)),
    Int[((a^2-b^2)*(n+1)-(a*b*(n+1))*sin[c+d*x]^(-1)+(b^2*(n+2))*sin[c+d*x]^(-2))*
      (a+b*sin[c+d*x]^(-1))^(n+1),x]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2-b^2] && RationalQ[n] && n<-1
```

- **Derivation: Rule 3a with** $m = 0$ **and** $k = -1$
- **Rule 18: If** $a^2 - b^2 \neq 0 \land n > 2$**, then**

$$\int (a + b \,\mathrm{Csc}[c + d\, x])^n \,dx \longrightarrow$$

$$-\frac{b^2 \,\mathrm{Cot}[c + d\, x] \,(a + b \,\mathrm{Csc}[c + d\, x])^{n-2}}{d\, (n - 1)} + \frac{1}{n - 1} \cdot$$

$$\int \left(a^3\, (n - 1) + b \left(b^2\, (n - 2) + 3\, a^2\, (n - 1)\right) \mathrm{Csc}[c + d\, x] + a\, b^2\, (3\, n - 4) \,\mathrm{Csc}[c + d\, x]^2\right) (a + b \,\mathrm{Csc}[c + d\, x])^{n-3} \,dx$$

- **Program code:**

```
Int[(a_+b_.*sin[c_.+d_.*x_]^(-1))^n_,x_Symbol] :=
  -b^2*Cot[c+d*x]*(a+b*Csc[c+d*x])^(n-2)/(d*(n-1)) +
  Dist[1/(n-1),
    Int[(a^3*(n-1)+(b*(b^2*(n-2)+3*a^2*(n-1)))*sin[c+d*x]^(-1)+(a*b^2*(3*n-4))*sin[c+d*x]^(-2))*
      (a+b*sin[c+d*x]^(-1))^(n-3),x]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2-b^2] && RationalQ[n] && n>2
```

Rules 15 – 16: $\int (b \operatorname{Sin}[c + d x]^k)^n \, dx$

- **Derivation: Rule 10a inverted**
- **Rule 15: If $k^2 = 1 \wedge n < -1$, then**

$$\int (b \operatorname{Sin}[c + d x]^k)^n \, dx \longrightarrow \frac{2 \operatorname{Cos}[c + d x] \, (b \operatorname{Sin}[c + d x]^k)^{n+1}}{b \, d \, (2 n + k + 1)} + \frac{(2 n + k + 3)}{b^2 \, (2 n + k + 1)} \int (b \operatorname{Sin}[c + d x]^k)^{n+2} \, dx$$

- **Program code:**

```
Int[(b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  2*Cos[c+d*x]*(b*Sin[c+d*x]^k)^(n+1)/(b*d*(2*n+k+1)) +
  Dist[(2*n+k+3)/(b^2*(2*n+k+1)),Int[(b*sin[c+d*x]^k)^(n+2),x]] /;
FreeQ[{b,c,d},x] && OneQ[k^2] && RationalQ[n] && n<-1
```

- **Derivation: Rule 3a or 3b with $m = 0$ and $a = 0$**
- **Rule 16: If $k^2 = 1 \wedge n > 1$, then**

$$\int (b \operatorname{Sin}[c + d x]^k)^n \, dx \longrightarrow -\frac{2 b \operatorname{Cos}[c + d x] \, (b \operatorname{Sin}[c + d x]^k)^{n-1}}{d \, (2 n + k - 1)} + \frac{b^2 \, (2 n + k - 3)}{2 n + k - 1} \int (b \operatorname{Sin}[c + d x]^k)^{n-2} \, dx$$

- **Program code:**

```
Int[(b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  -2*b*Cos[c+d*x]*(b*Sin[c+d*x]^k)^(n-1)/(d*(2*n+k-1)) +
  Dist[b^2*(2*n+k-3)/(2*n+k-1),Int[(b*sin[c+d*x]^k)^(n-2),x]] /;
FreeQ[{b,c,d},x] && OneQ[k^2] && RationalQ[n] && n>1
```

Integration Rules for

$$\int \left(\sin^j(z)\right)^m \left(a + b\sin^k(z)\right)^n \, dz \text{ when } j^2 = 1 \bigwedge k^2 = 1$$

Rule g: $\int \frac{1}{\text{Sin}[c + d\,x]\,(a + b\,\text{Sin}[c + d\,x])} \, dx$

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{1}{z\,(a+b\,z)} = \frac{1}{a\,z} - \frac{b}{a\,(a+b\,z)}$
- **Rule g:**

$$\int \frac{1}{\text{Sin}[c + d\,x]\,(a + b\,\text{Sin}[c + d\,x])} \, dx \longrightarrow \frac{1}{a} \int \frac{1}{\text{Sin}[c + d\,x]} \, dx - \frac{b}{a} \int \frac{1}{a + b\,\text{Sin}[c + d\,x]} \, dx$$

- **Program code:**

```
Int[1/(sin[c_.+d_.*x_]*(a_.+b_.*sin[c_.+d_.*x_])),x_Symbol] :=
  1/a*Int[1/Sin[c+d*x],x] - b/a*Int[1/(a+b*Sin[c+d*x]),x] /;
FreeQ[{a,b,c,d},x]
```

```
Int[1/((a_.+b_.*sin[c_.+d_.*x_])*(e_+f_.*sin[c_.+d_.*x_])),x_Symbol] :=
  b/(b*e-a*f)*Int[1/(a+b*sin[c+d*x]),x] -
  f/(b*e-a*f)*Int[1/(e+f*sin[c+d*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && NonzeroQ[b*e-a*f]
```

Rule h: $\int \frac{1}{(a+b\,\text{Sin}[c+d\,x])\sqrt{e+f\,\text{Sin}[c+d\,x]}}\,dx$

- **Basis:** $\partial_z \text{EllipticPi}\left[2m, \frac{z}{2}-\frac{\pi}{4}, 2n\right] = \frac{1}{2\,(1-m+m\,\text{Sin}[z])\sqrt{1-n+n\,\text{Sin}[z]}}$

- **Rule h1: If** $a^2-b^2 \neq 0 \wedge e^2-f^2 \neq 0 \wedge e+f>0$, **then**

$$\int \frac{1}{(a+b\,\text{Sin}[c+d\,x])\sqrt{e+f\,\text{Sin}[c+d\,x]}}\,dx \longrightarrow \frac{2}{d\,(a+b)\sqrt{e+f}}\,\text{EllipticPi}\left[\frac{2b}{a+b}, \frac{c+d\,x}{2}-\frac{\pi}{4}, \frac{2f}{e+f}\right]$$

- **Program code:**

```
Int[1/((a_.+b_.*sin[c_.+Pi/2+d_.*x_])*Sqrt[e_.+f_.*sin[c_.+Pi/2+d_.*x_]]),x_Symbol] :=
  2/(d*(a+b)*Rt[e+f,2])*EllipticPi[Sim[2*b/(a+b)],(c+d*x)/2,Sim[2*f/(e+f)]] /;
FreeQ[{a,b,c,d,e,f},x] && NonzeroQ[a^2-b^2] && NonzeroQ[e^2-f^2] && PositiveQ[e+f]
```

```
Int[1/((a_.+b_.*sin[c_.+d_.*x_])*Sqrt[e_.+f_.*sin[c_.+d_.*x_]]),x_Symbol] :=
  2/(d*(a+b)*Rt[e+f,2])*EllipticPi[Sim[2*b/(a+b)],(c+d*x)/2-Pi/4,Sim[2*f/(e+f)]] /;
FreeQ[{a,b,c,d,e,f},x] && NonzeroQ[a^2-b^2] && NonzeroQ[e^2-f^2] && PositiveQ[e+f]
```

- **Derivation: Piecewise constant extraction**

- **Basis:** $\partial_z \frac{\sqrt{\frac{e+f\,g[z]}{e+f}}}{\sqrt{e+f\,g[z]}} = 0$

- **Note: Since** $\frac{e}{e+f}+\frac{f}{e+f} = 1 > 0$, **rule h1 applies to the resulting integrand.**

- **Rule h2: If** $a^2-b^2 \neq 0 \wedge e^2-f^2 \neq 0 \wedge \neg\,(e+f>0)$, **then**

$$\int \frac{1}{(a+b\,\text{Sin}[c+d\,x])\sqrt{e+f\,\text{Sin}[c+d\,x]}}\,dx \longrightarrow$$

$$\frac{\sqrt{\frac{e+f\,\text{Sin}[c+d\,x]}{e+f}}}{\sqrt{e+f\,\text{Sin}[c+d\,x]}} \int \frac{1}{(a+b\,\text{Sin}[c+d\,x])\sqrt{\frac{e}{e+f}+\frac{f}{e+f}\,\text{Sin}[c+d\,x]}}\,dx$$

- **Program code:**

```
Int[1/((a_.+b_.*sin[c_.+d_.*x_])*Sqrt[e_.+f_.*sin[c_.+d_.*x_]]),x_Symbol] :=
  Sqrt[(e+f*Sin[c+d*x])/(e+f)]/Sqrt[e+f*Sin[c+d*x]]*
    Int[1/((a+b*sin[c+d*x])*Sqrt[e/(e+f)+f/(e+f)*sin[c+d*x]]),x] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2-b^2] && NonzeroQ[e^2-f^2] && Not[PositiveQ[e+f]]
```

Rule i: $\int \frac{\sqrt{a + b \,\mathrm{Sin}[c + d\,x]}}{e + f \,\mathrm{Sin}[c + d\,x]} \, dx$

- **Derivation: Algebraic expansion**

- **Basis:** $\frac{\sqrt{a+b\,z}}{e+f\,z} = \frac{b}{f\,\sqrt{a+b\,z}} + \frac{a\,f-b\,e}{f\,(e+f\,z)\,\sqrt{a+b\,z}}$

- **Rule i: If** $a^2 - b^2 \neq 0 \bigwedge e^2 - f^2 \neq 0 \bigwedge a\,f - b\,e \neq 0$**, then**

$$\int \frac{\sqrt{a + b \,\mathrm{Sin}[c + d\,x]}}{e + f \,\mathrm{Sin}[c + d\,x]} \, dx \longrightarrow$$
$$\frac{b}{f} \int \frac{1}{\sqrt{a + b \,\mathrm{Sin}[c + d\,x]}} \, dx + \frac{a\,f - b\,e}{f} \int \frac{1}{(e + f \,\mathrm{Sin}[c + d\,x]) \sqrt{a + b \,\mathrm{Sin}[c + d\,x]}} \, dx$$

- **Program code:**

```
Int[Sqrt[a_.+b_.*sin[c_.+d_.*x_]]/(e_.+f_.*sin[c_.+d_.*x_]),x_Symbol] :=
  b/f*Int[1/Sqrt[a+b*sin[c+d*x]],x] +
  (a*f-b*e)/f*Int[1/((e+f*sin[c+d*x])*Sqrt[a+b*sin[c+d*x]]),x] /;
FreeQ[{a,b,c,d,e,f},x] && NonzeroQ[a^2-b^2] && NonzeroQ[e^2-f^2] && NonzeroQ[a*f-b*e]
```

Rule j: $\int \frac{(a+b\,\mathrm{Sin}[c+d\,x])^{3/2}}{e+f\,\mathrm{Sin}[c+d\,x]}\,dx$

- **Derivation: Algebraic expansion**

- **Basis:** $\frac{(a+b\,z)^{3/2}}{e+f\,z}=\frac{b\sqrt{a+b\,z}}{f}+\frac{(a\,f-b\,e)\sqrt{a+b\,z}}{f\,(e+f\,z)}$

- **Rule j: If** $a^2-b^2\neq 0\bigwedge e^2-f^2\neq 0\bigwedge a\,f-b\,e\neq 0$, **then**

$$\int \frac{(a+b\,\mathrm{Sin}[c+d\,x])^{3/2}}{e+f\,\mathrm{Sin}[c+d\,x]}\,dx\longrightarrow \frac{b}{f}\int \sqrt{a+b\,\mathrm{Sin}[c+d\,x]}\,dx+\frac{a\,f-b\,e}{f}\int \frac{\sqrt{a+b\,\mathrm{Sin}[c+d\,x]}}{e+f\,\mathrm{Sin}[c+d\,x]}\,dx$$

- **Program code:**

```
Int[(a_.+b_.*sin[c_.+d_.*x_])^(3/2)/(e_.+f_.*sin[c_.+d_.*x_]),x_Symbol] :=
  b/f*Int[Sqrt[a+b*sin[c+d*x]],x] +
  Dist[(a*f-b*e)/f,Int[Sqrt[a+b*sin[c+d*x]]/(e+f*sin[c+d*x]),x]] /;
FreeQ[{a,b,c,d,e,f},x] && NonzeroQ[a^2-b^2] && NonzeroQ[e^2-f^2] && NonzeroQ[a*f-b*e]
```

$$\text{Rule k:}\quad \int \frac{1}{\sqrt{\text{Sin}[c+d\,x]}\,\sqrt{a+b\,\text{Sin}[c+d\,x]}}\,dx$$

- **Derivation: Algebraic expansion**

- **Basis: If** $b>0\wedge b-a>0$**, then** $\sqrt{a+b\,z}=\sqrt{1+z}\,\sqrt{\frac{a+b\,z}{1+z}}$

- **Rule k1: If** $b>0\wedge b^2-a^2>0$**, then**

$$\int \frac{1}{\sqrt{\text{Sin}[c+d\,x]}\,\sqrt{a+b\,\text{Sin}[c+d\,x]}}\,dx\ \longrightarrow\ \frac{2}{d\,\sqrt{a+b}}\,\text{EllipticF}\Big[\text{ArcSin}\Big[\text{Tan}\Big[\frac{c+d\,x}{2}-\frac{\pi}{4}\Big]\Big],\,-\frac{a-b}{a+b}\Big]$$

- **Program code:**

```
Int[1/(Sqrt[sin[c_.+d_.*x_]]*Sqrt[a_+b_.*sin[c_.+d_.*x_]]),x_Symbol] :=
  2/(d*Sqrt[a+b])*EllipticF[ArcSin[Tan[(c-Pi/2+d*x)/2]],-Sim[(a-b)/(a+b)]] /;
FreeQ[{a,b,c,d},x] && PositiveQ[b] && PositiveQ[b^2-a^2]
```

- **Derivation: Piecewise constant extraction**

- **Basis:** $\partial_z\left(\frac{\sqrt{1+f[z]}}{\sqrt{a+b\,f[z]}}\,\sqrt{\frac{a+b\,f[z]}{(a+b)\,(1+f[z])}}\right)=0$

- **Rule k2: If** $a^2-b^2\neq 0$**, then**

$$\int \frac{1}{\sqrt{\text{Sin}[c+d\,x]}\,\sqrt{a+b\,\text{Sin}[c+d\,x]}}\ \longrightarrow$$

$$\frac{2\,\sqrt{1+\text{Sin}[c+d\,x]}}{d\,\sqrt{a+b\,\text{Sin}[c+d\,x]}}\,\sqrt{\frac{a+b\,\text{Sin}[c+d\,x]}{(a+b)\,(1+\text{Sin}[c+d\,x])}}\ \text{EllipticF}\Big[\text{ArcSin}\Big[\text{Tan}\Big[\frac{c+d\,x}{2}-\frac{\pi}{4}\Big]\Big],\,-\frac{a-b}{a+b}\Big]$$

- **Program code:**

```
Int[1/(Sqrt[sin[c_.+d_.*x_]]*Sqrt[a_+b_.*sin[c_.+d_.*x_]]),x_Symbol] :=
  2*Sqrt[1+Sin[c+d*x]]/(d*Sqrt[a+b*Sin[c+d*x]])*
   Sqrt[(a+b*Sin[c+d*x])/((a+b)*(1+Sin[c+d*x]))]*
  EllipticF[ArcSin[Tan[(c-Pi/2+d*x)/2]],-Sim[(a-b)/(a+b)]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2-b^2]
```

Rule l: $\int \sqrt{\text{Sin}[c + d\,x]}\,\sqrt{a + b\,\text{Sin}[c + d\,x]}\,dx$

- **Derivation: Algebraic expansion**

- **Rule l: If $a^2 - b^2 \neq 0$, then**

$$\int \sqrt{\text{Sin}[c + d\,x]}\,\sqrt{a + b\,\text{Sin}[c + d\,x]}\,dx \longrightarrow$$

$$-a\int \frac{1}{\sqrt{\text{Sin}[c + d\,x]}\,\sqrt{a + b\,\text{Sin}[c + d\,x]}}\,dx + \int \frac{a + a\,\text{Sin}[c + d\,x] + b\,\text{Sin}[c + d\,x]^2}{\sqrt{\text{Sin}[c + d\,x]}\,\sqrt{a + b\,\text{Sin}[c + d\,x]}}\,dx$$

- **Program code:**

```
Int[Sqrt[sin[c_.+d_.*x_]]*Sqrt[a_+b_.*sin[c_.+d_.*x_]],x_Symbol] :=
  -a*Int[1/(Sqrt[sin[c+d*x]]*Sqrt[a+b*sin[c+d*x]]),x] +
  Int[(a+a*sin[c+d*x]+b*sin[c+d*x]^2)/(Sqrt[sin[c+d*x]]*Sqrt[a+b*sin[c+d*x]]),x] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2-b^2]
```

Rule m: $\int \frac{\sqrt{a + b \operatorname{Sin}[c + d x]}}{\sqrt{e + f \operatorname{Sin}[c + d x]}} \, dx$

- **Derivation: Algebraic expansion**
- **Rule m1: If $a^2 - b^2 \neq 0$, then**

$$\int \frac{\sqrt{\operatorname{Sin}[c + d x]}}{\sqrt{a + b \operatorname{Sin}[c + d x]}} \, dx \longrightarrow$$
$$-\int \frac{1}{\sqrt{\operatorname{Sin}[c + d x]} \sqrt{a + b \operatorname{Sin}[c + d x]}} \, dx + \int \frac{1 + \operatorname{Sin}[c + d x]}{\sqrt{\operatorname{Sin}[c + d x]} \sqrt{a + b \operatorname{Sin}[c + d x]}} \, dx$$

- **Program code:**

```
Int[Sqrt[sin[c_.+d_.*x_]]/Sqrt[a_+b_.*sin[c_.+d_.*x_]],x_Symbol] :=
  -Int[1/(Sqrt[sin[c+d*x]]*Sqrt[a+b*sin[c+d*x]]),x] +
  Int[(1+sin[c+d*x])/(Sqrt[sin[c+d*x]]*Sqrt[a+b*sin[c+d*x]]),x] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2-b^2]
```

- **Derivation: Algebraic expansion**
- **Rule m2: If $a^2 - b^2 \neq 0$, then**

$$\int \frac{\sqrt{a + b \operatorname{Sin}[c + d x]}}{\sqrt{\operatorname{Sin}[c + d x]}} \, dx \longrightarrow$$
$$(a - b) \int \frac{1}{\sqrt{\operatorname{Sin}[c + d x]} \sqrt{a + b \operatorname{Sin}[c + d x]}} \, dx + b \int \frac{1 + \operatorname{Sin}[c + d x]}{\sqrt{\operatorname{Sin}[c + d x]} \sqrt{a + b \operatorname{Sin}[c + d x]}} \, dx$$

- **Program code:**

```
Int[Sqrt[a_+b_.*sin[c_.+d_.*x_]]/Sqrt[sin[c_.+d_.*x_]],x_Symbol] :=
  (a-b)*Int[1/(Sqrt[sin[c+d*x]]*Sqrt[a+b*sin[c+d*x]]),x] +
  Dist[b,Int[(1+sin[c+d*x])/(Sqrt[sin[c+d*x]]*Sqrt[a+b*sin[c+d*x]]),x]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2-b^2]
```

- **Derivation: Algebraic expansion**
- **Note: This rule unifies rules m1 and m2, but requires messy application conditions.**
- **Rule: If $a^2 - b^2 \neq 0 \land e^2 - f^2 \neq 0 \land b e - a f \neq 0$, then**

$$\int \frac{\sqrt{a + b \operatorname{Sin}[c + d x]}}{\sqrt{e + f \operatorname{Sin}[c + d x]}} \, dx \longrightarrow$$

$$(a - b) \int \frac{1}{\sqrt{a + b \operatorname{Sin}[c + d x]} \sqrt{e + f \operatorname{Sin}[c + d x]}} \, dx + b \int \frac{1 + \operatorname{Sin}[c + d x]}{\sqrt{a + b \operatorname{Sin}[c + d x]} \sqrt{e + f \operatorname{Sin}[c + d x]}} \, dx$$

- **Program code:**

```
(* Int[Sqrt[a_.+b_.*sin[c_.+d_.*x_]]/Sqrt[e_.+f_.*sin[c_.+d_.*x_]],x_Symbol] :=
  (a-b)*Int[1/(Sqrt[a+b*sin[c+d*x]]*Sqrt[e+f*sin[c+d*x]]),x] +
  Dist[b,Int[(1+sin[c+d*x])/(Sqrt[a+b*sin[c+d*x]]*Sqrt[e+f*sin[c+d*x]]),x]] /;
FreeQ[{a,b,c,d,e,f},x] && NonzeroQ[a^2-b^2] && NonzeroQ[e^2-f^2] && NonzeroQ[b*e-a*f] *)
```

Rule n: $\int \frac{\sqrt{a+b\,\mathrm{Sin}[c+d\,x]}}{(e+f\,\mathrm{Sin}[c+d\,x])^{3/2}}\,dx$

- **Derivation: Algebraic expansion**

- **Rule n1: If $a^2-b^2\neq 0$, then**

$$\int \frac{\sqrt{\mathrm{Sin}[c+d\,x]}}{(a+b\,\mathrm{Sin}[c+d\,x])^{3/2}}\,dx \longrightarrow$$
$$-\frac{1}{a-b}\int \frac{1}{\sqrt{\mathrm{Sin}[c+d\,x]}\,\sqrt{a+b\,\mathrm{Sin}[c+d\,x]}}\,dx+\frac{a}{a-b}\int \frac{1+\mathrm{Sin}[c+d\,x]}{\sqrt{\mathrm{Sin}[c+d\,x]}\,(a+b\,\mathrm{Sin}[c+d\,x])^{3/2}}\,dx$$

- **Program code:**

```
Int[Sqrt[sin[c_.+d_.*x_]]/(a_+b_.*sin[c_.+d_.*x_])^(3/2),x_Symbol] :=
  -1/(a-b)*Int[1/(Sqrt[sin[c+d*x]]*Sqrt[a+b*sin[c+d*x]]),x] +
  Dist[a/(a-b),Int[(1+sin[c+d*x])/(Sqrt[sin[c+d*x]]*(a+b*sin[c+d*x])^(3/2)),x]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2-b^2]
```

- **Derivation: Algebraic expansion**

- **Rule n2: If $a^2-b^2\neq 0$, then**

$$\int \frac{\sqrt{a+b\,\mathrm{Sin}[c+d\,x]}}{\mathrm{Sin}[c+d\,x]^{3/2}}\,dx \longrightarrow$$
$$(a+b)\int \frac{1}{\sqrt{\mathrm{Sin}[c+d\,x]}\,\sqrt{a+b\,\mathrm{Sin}[c+d\,x]}}\,dx+a\int \frac{1-\mathrm{Sin}[c+d\,x]}{\mathrm{Sin}[c+d\,x]^{3/2}\,\sqrt{a+b\,\mathrm{Sin}[c+d\,x]}}\,dx$$

- **Program code:**

```
Int[Sqrt[a_+b_.*sin[c_.+d_.*x_]]/sin[c_.+d_.*x_]^(3/2),x_Symbol] :=
  Dist[a+b,Int[1/(Sqrt[sin[c+d*x]]*Sqrt[a+b*sin[c+d*x]]),x]] +
  Dist[a,Int[(1-sin[c+d*x])/(sin[c+d*x]^(3/2)*Sqrt[a+b*sin[c+d*x]]),x]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2-b^2]
```

- **Derivation: Algebraic expansion**
- **Note: This rule unifies rules n1 and n2, but requires messy application conditions.**
- **Rule: If $a^2 - b^2 \neq 0 \land e^2 - f^2 \neq 0 \land b e - a f \neq 0$, then**

$$\int \frac{\sqrt{a + b \operatorname{Sin}[c + d x]}}{(e + f \operatorname{Sin}[c + d x])^{3/2}} \, dx \longrightarrow$$

$$\frac{a - b}{e - f} \int \frac{1}{\sqrt{a + b \operatorname{Sin}[c + d x]} \sqrt{e + f \operatorname{Sin}[c + d x]}} \, dx + \frac{b e - a f}{e - f} \int \frac{1 + \operatorname{Sin}[c + d x]}{\sqrt{a + b \operatorname{Sin}[c + d x]} (e + f \operatorname{Sin}[c + d x])^{3/2}} \, dx$$

- **Program code:**

```
(* Int[Sqrt[a_.+b_.*sin[c_.+d_.*x_]]/(e_.+f_.*sin[c_.+d_.*x_])^(3/2),x_Symbol] :=
  Dist[(a-b)/(e-f),Int[1/(Sqrt[a+b*sin[c+d*x]]*Sqrt[e+f*sin[c+d*x]]),x]] +
  Dist[(b*e-a*f)/(e-f),Int[(1+sin[c+d*x])/(Sqrt[a+b*sin[c+d*x]]*(e+f*sin[c+d*x])^(3/2)),x]] /;
FreeQ[{a,b,c,d,e,f},x] && NonzeroQ[a^2-b^2] && NonzeroQ[e^2-f^2] && NonzeroQ[b*e-a*f] *)
```

Rule o: $\int \frac{(a+b\,\mathrm{Sin}[c+d\,x])^{3/2}}{\sqrt{e+f\,\mathrm{Sin}[c+d\,x]}}\,dx$

- **Derivation: Algebraic expansion**
- **Rule o1: If** $a^2-b^2\neq 0$**, then**

$$\int \frac{\mathrm{Sin}[c+d\,x]^{3/2}}{\sqrt{a+b\,\mathrm{Sin}[c+d\,x]}}\,dx \longrightarrow$$

$$-\frac{a}{2\,b}\int \frac{1+\mathrm{Sin}[c+d\,x]}{\sqrt{\mathrm{Sin}[c+d\,x]}\,\sqrt{a+b\,\mathrm{Sin}[c+d\,x]}}\,dx+\frac{1}{2\,b}\int \frac{a+a\,\mathrm{Sin}[c+d\,x]+2\,b\,\mathrm{Sin}[c+d\,x]^2}{\sqrt{\mathrm{Sin}[c+d\,x]}\,\sqrt{a+b\,\mathrm{Sin}[c+d\,x]}}\,dx$$

- **Program code:**

```
Int[sin[c_.+d_.*x_]^(3/2)/Sqrt[a_+b_.*sin[c_.+d_.*x_]],x_Symbol] :=
  -a/(2*b)*Int[(1+sin[c+d*x])/(Sqrt[sin[c+d*x]]*Sqrt[a+b*sin[c+d*x]]),x] +
  Dist[1/(2*b),
    Int[(a+a*sin[c+d*x]+2*b*sin[c+d*x]^2)/(Sqrt[sin[c+d*x]]*Sqrt[a+b*sin[c+d*x]]),x]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2-b^2]
```

- **Derivation: Algebraic expansion**
- **Rule o2: If** $a^2-b^2\neq 0$**, then**

$$\int \frac{(a+b\,\mathrm{Sin}[c+d\,x])^{3/2}}{\sqrt{\mathrm{Sin}[c+d\,x]}}\,dx \longrightarrow$$

$$\frac{3\,a\,b}{2}\int \frac{1+\mathrm{Sin}[c+d\,x]}{\sqrt{\mathrm{Sin}[c+d\,x]}\,\sqrt{a+b\,\mathrm{Sin}[c+d\,x]}}\,dx+\frac{1}{2}\int \frac{a\,(2\,a-3\,b)+a\,b\,\mathrm{Sin}[c+d\,x]+2\,b^2\,\mathrm{Sin}[c+d\,x]^2}{\sqrt{\mathrm{Sin}[c+d\,x]}\,\sqrt{a+b\,\mathrm{Sin}[c+d\,x]}}\,dx$$

- **Program code:**

```
Int[(a_+b_.*sin[c_.+d_.*x_])^(3/2)/Sqrt[sin[c_.+d_.*x_]],x_Symbol] :=
  3*a*b/2*Int[(1+sin[c+d*x])/(Sqrt[sin[c+d*x]]*Sqrt[a+b*sin[c+d*x]]),x] +
  Dist[1/2,
    Int[(a*(2*a-3*b)+a*b*sin[c+d*x]+2*b^2*sin[c+d*x]^2)/(Sqrt[sin[c+d*x]]*Sqrt[a+b*sin[c+d*x]]),x]]
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2-b^2]
```

Rule p: $\int \frac{(a+b\,\text{Sin}[c+d\,x])^{3/2}}{(e+f\,\text{Sin}[c+d\,x])^{3/2}}\,dx$

- **Derivation: Algebraic expansion**
- **Rule p1: If** $a^2-b^2\neq 0$**, then**

$$\int \frac{\text{Sin}[c+d\,x]^{3/2}}{(a+b\,\text{Sin}[c+d\,x])^{3/2}}\,dx \longrightarrow \frac{1}{b}\int \frac{1+\text{Sin}[c+d\,x]}{\sqrt{\text{Sin}[c+d\,x]}\,\sqrt{a+b\,\text{Sin}[c+d\,x]}}\,dx + \frac{1}{b}\int \frac{-a-(a+b)\,\text{Sin}[c+d\,x]}{\sqrt{\text{Sin}[c+d\,x]}\,(a+b\,\text{Sin}[c+d\,x])^{3/2}}\,dx$$

- **Program code:**

```
Int[sin[c_.+d_.*x_]^(3/2)/(a_+b_.*sin[c_.+d_.*x_])^(3/2),x_Symbol] :=
  1/b*Int[(1+sin[c+d*x])/(Sqrt[sin[c+d*x]]*Sqrt[a+b*sin[c+d*x]]),x] +
  Dist[1/b,Int[(-a-(a+b)*sin[c+d*x])/(Sqrt[sin[c+d*x]]*(a+b*sin[c+d*x])^(3/2)),x]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2-b^2]
```

- **Derivation: Algebraic expansion**
- **Rule p2: If** $a^2-b^2\neq 0$**, then**

$$\int \frac{(a+b\,\text{Sin}[c+d\,x])^{3/2}}{\text{Sin}[c+d\,x]^{3/2}}\,dx \longrightarrow b^2\int \frac{1+\text{Sin}[c+d\,x]}{\sqrt{\text{Sin}[c+d\,x]}\,\sqrt{a+b\,\text{Sin}[c+d\,x]}}\,dx + \int \frac{a^2+b\,(2\,a-b)\,\text{Sin}[c+d\,x]}{(\text{Sin}[c+d\,x])^{3/2}\,\sqrt{a+b\,\text{Sin}[c+d\,x]}}\,dx$$

- **Program code:**

```
Int[(a_+b_.*sin[c_.+d_.*x_])^(3/2)/sin[c_.+d_.*x_]^(3/2),x_Symbol] :=
  b^2*Int[(1+sin[c+d*x])/(Sqrt[sin[c+d*x]]*Sqrt[a+b*sin[c+d*x]]),x] +
  Int[(a^2+b*(2*a-b)*sin[c+d*x])/(sin[c+d*x]^(3/2)*Sqrt[a+b*sin[c+d*x]]),x] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2-b^2]
```

- **Derivation: Algebraic expansion**
- **Note: This rule unifies rules p1 and p2, but requires messy application conditions.**
- **Rule: If $a^2 - b^2 \neq 0 \land e^2 - f^2 \neq 0 \land b e - a f \neq 0$, then**

$$\int \frac{(a + b \operatorname{Sin}[c + d x])^{3/2}}{(e + f \operatorname{Sin}[c + d x])^{3/2}} \, dx \longrightarrow \frac{b^2}{f} \int \frac{1 + \operatorname{Sin}[c + d x]}{\sqrt{a + b \operatorname{Sin}[c + d x]} \, \sqrt{e + f \operatorname{Sin}[c + d x]}} \, dx +$$
$$\frac{1}{f} \int \frac{a^2 f - b^2 e + b \, (2 a f - b \, (e + f)) \operatorname{Sin}[c + d x]}{\sqrt{a + b \operatorname{Sin}[c + d x]} \, (e + f \operatorname{Sin}[c + d x])^{3/2}} \, dx$$

- **Program code:**

```
(* Int[(a_.+b_.*sin[c_.+d_.*x_])^(3/2)/(e_.+f_.*sin[c_.+d_.*x_])^(3/2),x_Symbol] :=
  b^2/f*Int[(1+sin[c+d*x])/(Sqrt[a+b*sin[c+d*x]]*Sqrt[e+f*sin[c+d*x]]),x] +
  Dist[1/f,
    Int[(a^2*f-b^2*e+b*(2*a*f-b*(e+f))*sin[c+d*x])/(Sqrt[a+b*sin[c+d*x]]*(e+f*sin[c+d*x])^(3/2)),x]]
FreeQ[{a,b,c,d,e,f},x] && NonzeroQ[a^2-b^2] && NonzeroQ[e^2-f^2] && NonzeroQ[b*e-a*f] *)
```

Rule q: $\int \frac{1}{\sqrt{a+b\,\text{Sin}[c+d\,x]}\,(e+f\,\text{Sin}[c+d\,x])^{3/2}}\,dx$

- **Derivation: Algebraic expansion**
- **Rule q1: If $a^2-b^2\neq 0$, then**

$$\int \frac{1}{\sqrt{\text{Sin}[c+d\,x]}\,(a+b\,\text{Sin}[c+d\,x])^{3/2}}\,dx \longrightarrow$$

$$\frac{1}{a-b}\int \frac{1}{\sqrt{\text{Sin}[c+d\,x]}\,\sqrt{a+b\,\text{Sin}[c+d\,x]}}\,dx-\frac{b}{a-b}\int \frac{1+\text{Sin}[c+d\,x]}{\sqrt{\text{Sin}[c+d\,x]}\,(a+b\,\text{Sin}[c+d\,x])^{3/2}}\,dx$$

- **Program code:**

```
Int[1/(Sqrt[sin[c_.+d_.*x_]]*(a_+b_.*sin[c_.+d_.*x_])^(3/2)),x_Symbol] :=
  1/(a-b)*Int[1/(Sqrt[sin[c+d*x]]*Sqrt[a+b*sin[c+d*x]]),x] -
  Dist[b/(a-b),Int[(1+sin[c+d*x])/(Sqrt[sin[c+d*x]]*(a+b*sin[c+d*x])^(3/2)),x]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2-b^2]
```

- **Derivation: Algebraic expansion**
- **Rule q2: If $a^2-b^2\neq 0$, then**

$$\int \frac{1}{\text{Sin}[c+d\,x]^{3/2}\,\sqrt{a+b\,\text{Sin}[c+d\,x]}}\,dx \longrightarrow$$

$$\int \frac{1}{\sqrt{\text{Sin}[c+d\,x]}\,\sqrt{a+b\,\text{Sin}[c+d\,x]}}\,dx+\int \frac{1-\text{Sin}[c+d\,x]}{\text{Sin}[c+d\,x]^{3/2}\,\sqrt{a+b\,\text{Sin}[c+d\,x]}}\,dx$$

- **Program code:**

```
Int[1/(sin[c_.+d_.*x_]^(3/2)*Sqrt[a_+b_.*sin[c_.+d_.*x_]]),x_Symbol] :=
  Int[1/(Sqrt[sin[c+d*x]]*Sqrt[a+b*sin[c+d*x]]),x] +
  Int[(1-sin[c+d*x])/(sin[c+d*x]^(3/2)*Sqrt[a+b*sin[c+d*x]]),x] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2-b^2]
```

$$\int \frac{\sqrt{a + b\,\text{Sin}[c + d\,x]}}{\sqrt{\text{Sin}[c + d\,x]}\,(A + A\,\text{Sin}[c + d\,x])}\,dx$$

- **Basis: If** $b > 0 \wedge b - a > 0$, **then** $\sqrt{a + b\,z} = \sqrt{1 + z}\,\sqrt{\frac{a+b\,z}{1+z}}$

- **Rule: If** $b > 0 \wedge b^2 - a^2 > 0$, **then**

$$\int \frac{\sqrt{a + b\,\text{Sin}[c + d\,x]}}{\sqrt{\text{Sin}[c + d\,x]}\,(A + A\,\text{Sin}[c + d\,x])}\,dx \longrightarrow \frac{\sqrt{a + b}}{d\,A}\,\text{EllipticE}\Big[\text{ArcSin}\Big[\text{Tan}\Big[\frac{c + d\,x}{2} - \frac{\pi}{4}\Big]\Big],\,-\frac{a - b}{a + b}\Big]$$

- **Program code:**

```
Int[Sqrt[a_+b_.*sin[c_.+d_.*x_]]/(Sqrt[sin[c_.+d_.*x_]]*(A_+B_.*sin[c_.+d_.*x_])),x_Symbol] :=
  Sqrt[a+b]/(d*A)*EllipticE[ArcSin[Tan[(c-Pi/2+d*x)/2]],-Sim[(a-b)/(a+b)]] /;
FreeQ[{a,b,c,d,A,B},x] && ZeroQ[A-B] && PositiveQ[b] && PositiveQ[b^2-a^2]
```

- **Derivation: Piecewise constant extraction**

- **Basis:** $\partial_z\left(\frac{\sqrt{1+f[z]}}{\sqrt{a+b\,f[z]}}\,\sqrt{\frac{a+b\,f[z]}{(a+b)\,(1+f[z])}}\right) = 0$

- **Rule: If** $a^2 - b^2 \neq 0$, **then**

$$\int \frac{\sqrt{a + b\,\text{Sin}[c + d\,x]}}{\sqrt{\text{Sin}[c + d\,x]}\,(A + A\,\text{Sin}[c + d\,x])} \longrightarrow$$

$$\frac{(a + b)\,\sqrt{1 + \text{Sin}[c + d\,x]}}{d\,A\,\sqrt{a + b\,\text{Sin}[c + d\,x]}}\,\sqrt{\frac{a + b\,\text{Sin}[c + d\,x]}{(a + b)\,(1 + \text{Sin}[c + d\,x])}}\,\text{EllipticE}\Big[\text{ArcSin}\Big[\text{Tan}\Big[\frac{c + d\,x}{2} - \frac{\pi}{4}\Big]\Big],\,-\frac{a - b}{a + b}\Big]$$

- **Program code:**

```
Int[Sqrt[a_+b_.*sin[c_.+d_.*x_]]/(Sqrt[sin[c_.+d_.*x_]]*(A_+B_.*sin[c_.+d_.*x_])),x_Symbol] :=
  (a+b)*Sqrt[1+Sin[c+d*x]]/(d*A*Sqrt[a+b*Sin[c+d*x]])*Sqrt[(a+b*Sin[c+d*x])/((a+b)*(1+Sin[c+d*x]))]*
  EllipticE[ArcSin[Tan[(c-Pi/2+d*x)/2]],-Sim[(a-b)/(a+b)]] /;
FreeQ[{a,b,c,d,A,B},x] && ZeroQ[A-B] && NonzeroQ[a^2-b^2]
```

$$\int \frac{\sqrt{\operatorname{Sin}[c+d\,x]}}{\sqrt{a+b\operatorname{Sin}[c+d\,x]}\,(A+A\operatorname{Sin}[c+d\,x])}\,dx$$

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{\sqrt{z}}{\sqrt{a+b\,z}\,(1+z)} = \frac{a}{(a-b)\sqrt{z}\,\sqrt{a+b\,z}} - \frac{\sqrt{a+b\,z}}{(a-b)\sqrt{z}\,(1+z)}$
- **Rule: If** $a^2-b^2\neq 0$**, then**

$$\int \frac{\sqrt{\operatorname{Sin}[c+d\,x]}}{\sqrt{a+b\operatorname{Sin}[c+d\,x]}\,(A+A\operatorname{Sin}[c+d\,x])}\,dx \longrightarrow$$

$$\frac{a}{A\,(a-b)}\int \frac{1}{\sqrt{\operatorname{Sin}[c+d\,x]}\,\sqrt{a+b\operatorname{Sin}[c+d\,x]}}\,dx - \frac{1}{a-b}\int \frac{\sqrt{a+b\operatorname{Sin}[c+d\,x]}}{\sqrt{\operatorname{Sin}[c+d\,x]}\,(A+A\operatorname{Sin}[c+d\,x])}\,dx$$

- **Program code:**

```
Int[Sqrt[sin[c_.+d_.*x_]]/(Sqrt[a_.+b_.*sin[c_.+d_.*x_]]*(A_+B_.*sin[c_.+d_.*x_])),x_Symbol] :=
  a/(A*(a-b))*Int[1/(Sqrt[sin[c+d*x]]*Sqrt[a+b*sin[c+d*x]]),x] -
  1/(a-b)*Int[Sqrt[a+b*sin[c+d*x]]/(Sqrt[sin[c+d*x]]*(A+B*sin[c+d*x])),x] /;
FreeQ[{a,b,c,d,A,B},x] && ZeroQ[A-B] && NonzeroQ[a^2-b^2]
```

$$\int \frac{A + A \operatorname{Sin}[c + d x]}{\sqrt{\operatorname{Sin}[c + d x]}\, \sqrt{a + b \operatorname{Sin}[c + d x]}} \, dx$$

- **Derivation: Algebraic expansion**

- **Basis: If** $b > 0 \wedge b - a > 0$, **then** $\sqrt{a + b z} = \sqrt{1 + z} \sqrt{\frac{a+b z}{1+z}}$

- **Rule: If** $b > 0 \wedge b^2 - a^2 > 0$, **then**

$$\int \frac{A + A \operatorname{Sin}[c + d x]}{\sqrt{\operatorname{Sin}[c + d x]}\, \sqrt{a + b \operatorname{Sin}[c + d x]}} \, dx \longrightarrow \frac{4 A}{d \sqrt{a + b}} \operatorname{EllipticPi}\left[-1, \operatorname{ArcSin}\left[\operatorname{Tan}\left[\frac{c + d x}{2} - \frac{\pi}{4}\right]\right], -\frac{a - b}{a + b}\right]$$

- **Program code:**

```
Int[(A_+B_.*sin[c_.+d_.*x_])/(Sqrt[sin[c_.+d_.*x_]]*Sqrt[a_+b_.*sin[c_.+d_.*x_]]),x_Symbol] :=
  4*A/(d*Sqrt[a+b])*EllipticPi[-1,ArcSin[Tan[(c-Pi/2+d*x)/2]],-Sim[(a-b)/(a+b)]] /;
FreeQ[{a,b,c,d,A,B},x] && ZeroQ[A-B] && PositiveQ[b] && PositiveQ[b^2-a^2]
```

- **Derivation: Piecewise constant extraction**

- **Basis:** $\partial_z \left(\frac{\sqrt{1+f[z]}}{\sqrt{a+b f[z]}} \sqrt{\frac{a+b f[z]}{(a+b)(1+f[z])}} \right) = 0$

- **Rule: If** $a^2 - b^2 \neq 0$, **then**

$$\int \frac{A + A \operatorname{Sin}[c + d x]}{\sqrt{\operatorname{Sin}[c + d x]}\, \sqrt{a + b \operatorname{Sin}[c + d x]}} \longrightarrow$$

$$\frac{4 A \sqrt{1 + \operatorname{Sin}[c + d x]}}{d \sqrt{a + b \operatorname{Sin}[c + d x]}} \sqrt{\frac{a + b \operatorname{Sin}[c + d x]}{(a + b)(1 + \operatorname{Sin}[c + d x])}} \operatorname{EllipticPi}\left[-1, \operatorname{ArcSin}\left[\operatorname{Tan}\left[\frac{c + d x}{2} - \frac{\pi}{4}\right]\right], -\frac{a - b}{a + b}\right]$$

- **Program code:**

```
Int[(A_+B_.*sin[c_.+d_.*x_])/(Sqrt[sin[c_.+d_.*x_]]*Sqrt[a_+b_.*sin[c_.+d_.*x_]]),x_Symbol] :=
  4*A*Sqrt[1+Sin[c+d*x]]/(d*Sqrt[a+b*Sin[c+d*x]])*
   Sqrt[(a+b*Sin[c+d*x])/((a+b)*(1+Sin[c+d*x]))]*
  EllipticPi[-1,ArcSin[Tan[(c-Pi/2+d*x)/2]],-Sim[(a-b)/(a+b)]] /;
FreeQ[{a,b,c,d,A,B},x] && ZeroQ[A-B] && NonzeroQ[a^2-b^2]
```

Rule r: $\int (\text{Sin}[c+d\,x]^j)^m\,(b\,\text{Sin}[c+d\,x]^k)^n\,dx$

- **Derivation: Algebraic simplification**
- **Rule r1: If** $k^2 = 1 \wedge m \in \mathbb{Z}$, **then**

$$\int \text{Sin}[c+d\,x]^m\,(b\,\text{Sin}[c+d\,x]^k)^n\,dx \longrightarrow \frac{1}{b^{k\,m}}\int (b\,\text{Sin}[c+d\,x]^k)^{k\,m+n}\,dx$$

- **Program code:**

```
Int[sin[c_.+d_.*x_]^m_.*(b_.*sin[c_.+d_.*x_]^k_.)^n_.,x_Symbol] :=
  Dist[1/b^(k*m),Int[(b*sin[c+d*x]^k)^(k*m+n),x]] /;
FreeQ[{b,c,d,n},x] && OneQ[k^2] && IntegerQ[m]
```

- **Derivation: Piecewise constant extraction**
- **Basis: If** $j^2 = 1$, **then** $\partial_z \frac{\sqrt{b\,f[z]^k}}{\left(\sqrt{f[z]^j}\right)^{j\,k}} = 0$
- **Rule r2: If** $j^2 = k^2 = 1 \bigwedge m - \frac{1}{2} \in \mathbb{Z} \bigwedge n - \frac{1}{2} \in \mathbb{Z} \bigwedge n > 0$, **then**

$$\int (\text{Sin}[c+d\,x]^j)^m\,(b\,\text{Sin}[c+d\,x]^k)^n\,dx \longrightarrow \frac{b^{n-\frac{1}{2}}\sqrt{b\,\text{Sin}[c+d\,x]^k}}{\left(\sqrt{\text{Sin}[c+d\,x]^j}\right)^{j\,k}}\int \text{Sin}[c+d\,x]^{j\,m+k\,n}\,dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_*(b_*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  Dist[b^(n-1/2)*Sqrt[b*Sin[c+d*x]^k]/(Sqrt[Sin[c+d*x]^j])^(j*k),Int[sin[c+d*x]^(j*m+k*n),x]] /;
FreeQ[{b,c,d},x] && OneQ[j^2,k^2] && IntegerQ[m-1/2] && IntegerQ[n-1/2] && n>0
```

- **Derivation: Piecewise constant extraction**

- **Basis: If** $j^2 = 1$**, then** $\partial_z \frac{\left(\sqrt{f[z]^j}\right)^{j k}}{\sqrt{b f[z]^k}} = 0$

- **Rule r3: If** $j^2 = k^2 = 1 \bigwedge m - \frac{1}{2} \in \mathbb{Z} \bigwedge n - \frac{1}{2} \in \mathbb{Z} \bigwedge n < 0$**, then**

$$\int \left(\text{Sin}[c+d\,x]^j\right)^m \left(b\,\text{Sin}[c+d\,x]^k\right)^n \,dx \longrightarrow \frac{b^{n+\frac{1}{2}} \left(\sqrt{\text{Sin}[c+d\,x]^j}\right)^{j\,k}}{\sqrt{b\,\text{Sin}[c+d\,x]^k}} \int \text{Sin}[c+d\,x]^{j\,m+k\,n} \,dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_*(b_*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  Dist[b^(n+1/2)*(Sqrt[Sin[c+d*x]^j])^(j*k)/Sqrt[b*Sin[c+d*x]^k],Int[sin[c+d*x]^(j*m+k*n),x]] /;
FreeQ[{b,c,d},x] && OneQ[j^2,k^2] && IntegerQ[m-1/2] && IntegerQ[n-1/2] && n<0
```

Rules: $\int \left(\text{Sin}[c+d\,x]^j\right)^m (a+b\,\text{Csc}[c+d\,x])^n\,dx$

- **Derivation: Algebraic simplification**

- **Rule s1: If** $j^2 = 1 \bigwedge a^2 - b^2 \neq 0 \bigwedge -\frac{1}{2} \leq m + j \leq \frac{3}{2}$**, then**

$$\int \frac{\left(\text{Sin}[c+d\,x]^j\right)^m}{a+b\,\text{Csc}[c+d\,x]}\,dx \longrightarrow \int \frac{\left(\text{Sin}[c+d\,x]^j\right)^{m+j}}{b+a\,\text{Sin}[c+d\,x]}\,dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_./(a_+b_.*sin[c_.+d_.*x_]^(-1)),x_Symbol] :=
  Int[(sin[c+d*x]^j)^(m+j)/(b+a*sin[c+d*x]),x] /;
FreeQ[{a,b,c,d},x] && OneQ[j^2] && NonzeroQ[a^2-b^2] && RationalQ[m] && -1/2≤m+j≤3/2
```

- **Derivation: Piecewise constant extraction**

- **Basis:** $\partial_z \frac{\sqrt{b+a\,f[z]}}{\sqrt{f[z]}\,\sqrt{a+b/f[z]}} = 0$

- **Rule s2: If**
$a^2 - b^2 \neq 0 \bigwedge m \in \mathbb{Z} \bigwedge n - \frac{1}{2} \in \mathbb{Z} \bigwedge ((m = 1 \wedge -1 < n < 2) \vee (m = -1 \wedge -2 < n < 1) \vee (m = -2 \wedge -2 < n < 0))$**, then**

$$\int \text{Sin}[c+d\,x]^m (a+b\,\text{Csc}[c+d\,x])^n\,dx \longrightarrow$$
$$\frac{\sqrt{b+a\,\text{Sin}[c+d\,x]}}{\sqrt{\text{Sin}[c+d\,x]}\,\sqrt{a+b\,\text{Csc}[c+d\,x]}} \int \text{Sin}[c+d\,x]^{m-n} (b+a\,\text{Sin}[c+d\,x])^n\,dx$$

- **Program code:**

```
Int[sin[c_.+d_.*x_]^m_.*(a_+b_.*sin[c_.+d_.*x_]^(-1))^n_,x_Symbol] :=
  Dist[Sqrt[b+a*Sin[c+d*x]]/(Sqrt[Sin[c+d*x]]*Sqrt[a+b*Csc[c+d*x]]),
    Int[sin[c+d*x]^(m-n)*(b+a*sin[c+d*x])^n,x]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2-b^2] && IntegerQ[m] && IntegerQ[n-1/2] &&
  (m==1 && -1<n<2 || m==-1 && -2<n<1  || m==-2 && -2<n<0)
```

- **Derivation: Piecewise constant extraction**
- **Basis: If** $j^2 = 1$**, then** $\partial_z \frac{\sqrt{b+a\,f[z]}}{\left(\sqrt{f[z]^j}\right)^j \sqrt{a+b/f[z]}} = 0$
- **Rule s3: If** $j^2 = 1 \bigwedge a^2 - b^2 \neq 0 \bigwedge m - \frac{1}{2} \in \mathbb{Z} \bigwedge n - \frac{1}{2} \in \mathbb{Z} \bigwedge -1 \leq j\,m - n \leq 1$**, then**

$$\int \left(\text{Sin}[c+d\,x]^j\right)^m (a+b\,\text{Csc}[c+d\,x])^n \,dx \longrightarrow$$

$$\frac{\sqrt{b+a\,\text{Sin}[c+d\,x]}}{\sqrt{\text{Sin}[c+d\,x]^j}^{\,j} \sqrt{a+b\,\text{Csc}[c+d\,x]}} \int \text{Sin}[c+d\,x]^{j\,m-n} (b+a\,\text{Sin}[c+d\,x])^n \,dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_*(a_+b_.*sin[c_.+d_.*x_]^(-1))^n_,x_Symbol] :=
  Dist[Sqrt[b+a*Sin[c+d*x]]/((Sqrt[Sin[c+d*x]^j])^j*Sqrt[a+b*Csc[c+d*x]]),
    Int[sin[c+d*x]^(j*m-n)*(b+a*sin[c+d*x])^n,x]] /;
FreeQ[{a,b,c,d},x] && OneQ[j^2] && NonzeroQ[a^2-b^2] && IntegerQ[m-1/2] && IntegerQ[n-1/2] &&
  -1<n<2 && -1≤j*m-n≤1
```

Rule t: $\int \text{Csc}[c+d\,x]^{m/2}\left(a+b\,\text{Sin}[c+d\,x]^k\right)^n dx$

- **Derivation: Piecewise constant extraction**
- **Basis:** $\partial_z\left(\sqrt{f[z]}\,\sqrt{1/f[z]}\right) = 0$
- **Rule t: If** $k^2 = 1 \bigwedge a^2 - b^2 \neq 0 \bigwedge m - \frac{1}{2} \in \mathbb{Z} \bigwedge -1 \leq n < 1$**, then**

$$\int \text{Csc}[c+d\,x]^m\left(a+b\,\text{Sin}[c+d\,x]^k\right)^n dx \longrightarrow \sqrt{\text{Csc}[c+d\,x]}\,\sqrt{\text{Sin}[c+d\,x]}\int\frac{\left(a+b\,\text{Sin}[c+d\,x]^k\right)^n}{\text{Sin}[c+d\,x]^m}\,dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^(-1))^m_*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  Dist[Sqrt[Csc[c+d*x]]*Sqrt[Sin[c+d*x]],
    Int[(a+b*sin[c+d*x]^k)^n/sin[c+d*x]^m,x]] /;
FreeQ[{a,b,c,d},x] && OneQ[k^2] && NonzeroQ[a^2-b^2] && IntegerQ[m-1/2] && RationalQ[n] &&
  (k===1 || -1<m<1 && -1≤n<1)
```

Rules 11 – 12: $\int \left(\text{Sin}[c+d\,x]^j\right)^m \left(a+b\,\text{Sin}[c+d\,x]^k\right)^2 \,dx$

- **Derivation: Rule 4a with $n = 2$**

- **Rule 11: If $j^2 = k^2 = 1 \bigwedge j\,k\,m + \frac{k+1}{2} \neq 0 \bigwedge j\,k\,m \leq -1$, then**

$$\int \left(\text{Sin}[c+d\,x]^j\right)^m \left(a+b\,\text{Sin}[c+d\,x]^k\right)^2 \,dx \longrightarrow \frac{a^2\,\text{Cos}[c+d\,x]\,\left(\text{Sin}[c+d\,x]^j\right)^{m+j\,k}}{d\,\left(j\,k\,m+\frac{k+1}{2}\right)} +$$
$$\frac{1}{j\,k\,m+\frac{k+1}{2}} \int \left(\text{Sin}[c+d\,x]^j\right)^{m+j\,k} \left(2\,a\,b\,\left(j\,k\,m+\frac{k+1}{2}\right) + \left(a^2+\left(a^2+b^2\right)\left(j\,k\,m+\frac{k+1}{2}\right)\right) \text{Sin}[c+d\,x]^k\right) dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(a_+b_.*sin[c_.+d_.*x_]^k_.)^2,x_Symbol] :=
  a^2*Cos[c+d*x]*(Sin[c+d*x]^j)^(m+j*k)/(d*(j*k*m+(k+1)/2)) +
  Dist[1/(j*k*m+(k+1)/2),
    Int[(sin[c+d*x]^j)^(m+j*k)*
      Sim[2*a*b*(j*k*m+(k+1)/2)+(a^2+(a^2+b^2)*(j*k*m+(k+1)/2))*sin[c+d*x]^k,x],x]] /;
FreeQ[{a,b,c,d},x] && OneQ[j^2,k^2] && RationalQ[m] && j*k*m+(k+1)/2≠0 && j*k*m≤-1
```

- **Derivation: Rule 3a with $n = 2$**

- **Rule 12: If $j^2 = k^2 = 1 \bigwedge j\,k\,m + \frac{k+3}{2} \neq 0 \bigwedge j\,k\,m \geq -1$, then**

$$\int \left(\text{Sin}[c+d\,x]^j\right)^m \left(a+b\,\text{Sin}[c+d\,x]^k\right)^2 \,dx \longrightarrow -\frac{b^2\,\text{Cos}[c+d\,x]\,\left(\text{Sin}[c+d\,x]^j\right)^{m+j\,k}}{d\,\left(j\,k\,m+\frac{k+3}{2}\right)} +$$
$$\frac{1}{j\,k\,m+\frac{k+3}{2}} \int \left(\text{Sin}[c+d\,x]^j\right)^{m} \left(a^2+\left(a^2+b^2\right)\left(j\,k\,m+\frac{k+1}{2}\right) + 2\,a\,b\,\left(j\,k\,m+\frac{k+3}{2}\right) \text{Sin}[c+d\,x]^k\right) dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(a_+b_.*sin[c_.+d_.*x_]^k_.)^2,x_Symbol] :=
  -b^2*Cos[c+d*x]*(Sin[c+d*x]^j)^(m+j*k)/(d*(j*k*m+(k+3)/2)) +
  Dist[1/(j*k*m+(k+3)/2),
    Int[(sin[c+d*x]^j)^m*
      Sim[a^2+(a^2+b^2)*(j*k*m+(k+1)/2)+2*a*b*(j*k*m+(k+3)/2)*sin[c+d*x]^k,x],x]] /;
FreeQ[{a,b,c,d},x] && OneQ[j^2,k^2] && RationalQ[m] && j*k*m+(k+3)/2≠0 && j*k*m≥-1
```

Rules 9 – 10: $\int \mathrm{Sin}[c+d\,x]^{\frac{k-1}{2}} \left(a+b\,\mathrm{Sin}[c+d\,x]^{k}\right)^{n} \,dx$

- **Reference: G&R 2.552.3**

- **Derivation: Rule 1c with** $j\,m = \frac{k-1}{2}$

- **Rule 9: If** $k^2 = 1 \wedge a^2 - b^2 \neq 0 \wedge n < -1$, **then**

$$\int \mathrm{Sin}[c+d\,x]^{\frac{k-1}{2}} \left(a+b\,\mathrm{Sin}[c+d\,x]^{k}\right)^{n} dx \longrightarrow -\frac{b\,\mathrm{Cos}[c+d\,x]\,\mathrm{Sin}[c+d\,x]^{\frac{k-1}{2}} \left(a+b\,\mathrm{Sin}[c+d\,x]^{k}\right)^{n+1}}{d\,(n+1)\,\left(a^2-b^2\right)} +$$
$$\frac{1}{(n+1)\,\left(a^2-b^2\right)} \int \mathrm{Sin}[c+d\,x]^{\frac{k-1}{2}} \left(a\,(n+1) - b\,(n+2)\,\mathrm{Sin}[c+d\,x]^{k}\right) \left(a+b\,\mathrm{Sin}[c+d\,x]^{k}\right)^{n+1} dx$$

- **Program code:**

```
Int[(a_+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  -b*Cos[c+d*x]*(a+b*Sin[c+d*x])^(n+1)/(d*(n+1)*(a^2-b^2)) +
  Dist[1/((n+1)*(a^2-b^2)),Int[(a*(n+1)-b*(n+2)*sin[c+d*x])*(a+b*sin[c+d*x])^(n+1),x]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2-b^2] && RationalQ[n] && n<-1
```

```
Int[sin[c_.+d_.*x_]^(-1)*(a_+b_.*sin[c_.+d_.*x_]^(-1))^n_,x_Symbol] :=
  -b*Cot[c+d*x]*(a+b*Csc[c+d*x])^(n+1)/(d*(n+1)*(a^2-b^2)) +
  Dist[1/((n+1)*(a^2-b^2)),Int[sin[c+d*x]^(-1)*(a*(n+1)-b*(n+2)*sin[c+d*x]^(-1))*(a+b*sin[c+d*x]^(-1
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2-b^2] && RationalQ[n] && n<-1
```

- **Reference: G&R 2.552.3 inverted**

- **Derivation: Rule 3b with** $j\,m = \frac{k-1}{2}$

- **Rule 10: If** $k^2 = 1 \wedge a^2 - b^2 \neq 0 \wedge n > 1$, **then**

$$\int \mathrm{Sin}[c+d\,x]^{\frac{k-1}{2}} \left(a+b\,\mathrm{Sin}[c+d\,x]^{k}\right)^{n} dx \longrightarrow -\frac{b\,\mathrm{Cos}[c+d\,x]\,\mathrm{Sin}[c+d\,x]^{\frac{k-1}{2}} \left(a+b\,\mathrm{Sin}[c+d\,x]^{k}\right)^{n-1}}{d\,n} +$$
$$\frac{1}{n} \int (\mathrm{Sin}[c+d\,x])^{\frac{k-1}{2}} \left(a^2\,n + b^2\,(n-1) + a\,b\,(2\,n-1)\,\mathrm{Sin}[c+d\,x]^{k}\right) \left(a+b\,\mathrm{Sin}[c+d\,x]^{k}\right)^{n-2} dx$$

- **Program code:**

```
Int[(a_.+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  -b*Cos[c+d*x]*(a+b*Sin[c+d*x])^(n-1)/(d*n) +
  Dist[1/n,Int[Sim[a^2*n+b^2*(n-1)+a*b*(2*n-1)*sin[c+d*x],x]*(a+b*sin[c+d*x])^(n-2),x]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2-b^2] && RationalQ[n] && n>1
```

```
Int[sin[c_.+d_.*x_]^(-1)*(a_+b_.*sin[c_.+d_.*x_]^(-1))^n_,x_Symbol] :=
  -b*Cot[c+d*x]*(a+b*Csc[c+d*x])^(n-1)/(d*n) +
  Dist[1/n,Int[sin[c+d*x]^(-1)*(a^2*n+b^2*(n-1)+a*b*(2*n-1)*sin[c+d*x]^(-1))*(a+b*sin[c+d*x]^(-1))^(
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2-b^2] && RationalQ[n] && n>1
```

Rules 13 – 14: $\int \text{Sin}[c+d\,x]^{\frac{3k-1}{2}} \left(a+b\,\text{Sin}[c+d\,x]^k\right)^n dx$

- **Derivation: Rule 1b with** $j\,m = \frac{3k-1}{2}$
- **Rule 13: If** $k^2 = 1 \land a^2-b^2 \neq 0 \land n < -1$**, then**

$$\int \text{Sin}[c+d\,x]^{\frac{3k-1}{2}} \left(a+b\,\text{Sin}[c+d\,x]^k\right)^n dx \longrightarrow$$

$$\frac{a\,\text{Cos}[c+d\,x]\,\text{Sin}[c+d\,x]^{\frac{k-1}{2}} \left(a+b\,\text{Sin}[c+d\,x]^k\right)^{n+1}}{d\,(n+1)\left(a^2-b^2\right)} -$$

$$\frac{1}{(n+1)\left(a^2-b^2\right)} \int \text{Sin}[c+d\,x]^{\frac{k-1}{2}} \left(b\,(n+1)-a\,(n+2)\,\text{Sin}[c+d\,x]^k\right) \left(a+b\,\text{Sin}[c+d\,x]^k\right)^{n+1} dx$$

- **Program code:**

```
Int[sin[c_.+d_.*x_]^m_*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  a*Cos[c+d*x]*Sin[c+d*x]^((k-1)/2)*(a+b*Sin[c+d*x]^k)^(n+1)/(d*(n+1)*(a^2-b^2)) -
  Dist[1/((n+1)*(a^2-b^2)),
    Int[Sin[c+d*x]^((k-1)/2)*(b*(n+1)-a*(n+2)*Sin[c+d*x]^k)*(a+b*Sin[c+d*x]^k)^(n+1),x]] /;
FreeQ[{a,b,c,d},x] && OneQ[k^2] && ZeroQ[m-(3*k-1)/2] && NonzeroQ[a^2-b^2] && RationalQ[n] && n<-1
```

- **Derivation: Algebraic expansion**
- **Rule 14a: If** $k^2 = 1 \land a^2-b^2 \neq 0$**, then**

$$\int \frac{\text{Sin}[c+d\,x]^{\frac{3k-1}{2}}}{a+b\,\text{Sin}[c+d\,x]^k} dx \longrightarrow \frac{1}{b} \int \text{Sin}[c+d\,x]^{\frac{k-1}{2}} dx - \frac{a}{b} \int \frac{\text{Sin}[c+d\,x]^{\frac{k-1}{2}}}{a+b\,\text{Sin}[c+d\,x]^k} dx$$

- **Program code:**

```
Int[sin[c_.+d_.*x_]^m_/(a_+b_.*sin[c_.+d_.*x_]^k_.),x_Symbol] :=
  1/b*Int[Sin[c+d*x]^((k-1)/2),x] -
  a/b*Int[Sin[c+d*x]^((k-1)/2)/(a+b*Sin[c+d*x]^k),x] /;
FreeQ[{a,b,c,d},x] && OneQ[k^2] && ZeroQ[m-(3*k-1)/2] && NonzeroQ[a^2-b^2]
```

- **Derivation: Rule 2b with** $j\ m = \frac{3k-1}{2}$
- **Rule 14b: If** $k^2 = 1 \land n > 1$**, then**

$$\int \text{Sin}[c+d\,x]^{\frac{3k-1}{2}} \left(a+b\,\text{Sin}[c+d\,x]^k\right)^n dx \longrightarrow$$

$$-\frac{\text{Cos}[c+d\,x]\,\text{Sin}[c+d\,x]^{\frac{k-1}{2}} \left(a+b\,\text{Sin}[c+d\,x]^k\right)^n}{d\,(n+1)} +$$

$$\frac{n}{n+1}\int \text{Sin}[c+d\,x]^{\frac{k-1}{2}} \left(b+a\,\text{Sin}[c+d\,x]^k\right) \left(a+b\,\text{Sin}[c+d\,x]^k\right)^{n-1} dx$$

- **Program code:**

```
Int[sin[c_.+d_.*x_]^m_*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  -Cos[c+d*x]*Sin[c+d*x]^((k-1)/2)*(a+b*Sin[c+d*x]^k)^n/(d*(n+1)) +
  Dist[n/(n+1),
    Int[Sin[c+d*x]^((k-1)/2)*(b+a*Sin[c+d*x]^k)*(a+b*Sin[c+d*x]^k)^(n-1),x]] /;
FreeQ[{a,b,c,d},x] && OneQ[k^2] && ZeroQ[m-(3*k-1)/2] && RationalQ[n] && n>-1
```

Rules 20 – 21: $\int \mathrm{Sin}[c+d\,x]^{\frac{5k-1}{2}} \left(a+b\,\mathrm{Sin}[c+d\,x]^k\right)^n \,dx$

- **Derivation: Rule 1a with** $j\,m = \frac{5k-1}{2}$

- **Rule 20: If** $k^2 = 1 \land a^2 - b^2 \neq 0 \land n < -1$**, then**

$$\int \mathrm{Sin}[c+d\,x]^{\frac{5k-1}{2}} \left(a+b\,\mathrm{Sin}[c+d\,x]^k\right)^n \,dx \longrightarrow$$

$$-\frac{a^2\,\mathrm{Cos}[c+d\,x]\,\mathrm{Sin}[c+d\,x]^{\frac{k-1}{2}} \left(a+b\,\mathrm{Sin}[c+d\,x]^k\right)^{n+1}}{b\,d\,(n+1)\left(a^2-b^2\right)} +$$

$$\frac{1}{b\,(n+1)\left(a^2-b^2\right)} \int \mathrm{Sin}[c+d\,x]^{\frac{k-1}{2}} \left(a\,b\,(n+1) - \left(a^2+b^2\,(n+1)\right)\mathrm{Sin}[c+d\,x]^k\right) \left(a+b\,\mathrm{Sin}[c+d\,x]^k\right)^{n+1} dx$$

- **Program code:**

```
Int[sin[c_.+d_.*x_]^m_*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  -a^2*Cos[c+d*x]*Sin[c+d*x]^((k-1)/2)*(a+b*Sin[c+d*x]^k)^(n+1)/(b*d*(n+1)*(a^2-b^2)) +
  Dist[1/(b*(n+1)*(a^2-b^2)),
     Int[Sin[c+d*x]^((k-1)/2)*(a*b*(n+1)-(a^2+b^2*(n+1))*Sin[c+d*x]^k)*(a+b*Sin[c+d*x]^k)^(n+1),x]] /
FreeQ[{a,b,c,d},x] && OneQ[k^2] && ZeroQ[m-(5*k-1)/2] && NonzeroQ[a^2-b^2] && RationalQ[n] && n<-1
```

- **Derivation: Rule 21b with** $n = -1$

- **Rule 21a: If** $k^2 = 1$**, then**

$$\int \frac{\mathrm{Sin}[c+d\,x]^{\frac{5k-1}{2}}}{a+b\,\mathrm{Sin}[c+d\,x]^k} dx \longrightarrow -\frac{\mathrm{Cos}[c+d\,x]\,\mathrm{Sin}[c+d\,x]^{\frac{k-1}{2}}}{b\,d} - \frac{a}{b}\int \frac{\mathrm{Sin}[c+d\,x]^{\frac{3k-1}{2}}}{a+b\,\mathrm{Sin}[c+d\,x]^k} dx$$

- **Program code:**

```
Int[sin[c_.+d_.*x_]^m_/(a_+b_.*sin[c_.+d_.*x_]^k_.),x_Symbol] :=
  -Cos[c+d*x]*Sin[c+d*x]^((k-1)/2)/(b*d) -
   Dist[a/b,Int[Sin[c+d*x]^((3*k-1)/2)/(a+b*Sin[c+d*x]^k),x]] /;
FreeQ[{a,b,c,d},x] && OneQ[k^2] && ZeroQ[m-(5*k-1)/2]
```

- **Derivation: Rule 2a with** $j\,m = \frac{5k-1}{2}$
- **Rule 21b: If** $k^2 = 1 \land n > 1$**, then**

$$\int \mathrm{Sin}[c+d\,x]^{\frac{5k-1}{2}} \left(a+b\,\mathrm{Sin}[c+d\,x]^k\right)^n dx \longrightarrow$$

$$-\frac{\mathrm{Cos}[c+d\,x]\,\mathrm{Sin}[c+d\,x]^{\frac{k-1}{2}} \left(a+b\,\mathrm{Sin}[c+d\,x]^k\right)^{n+1}}{b\,d\,(n+2)} +$$

$$\frac{1}{b\,(n+2)} \int \mathrm{Sin}[c+d\,x]^{\frac{k-1}{2}} \left(b\,(n+1) - a\,\mathrm{Sin}[c+d\,x]^k\right) \left(a+b\,\mathrm{Sin}[c+d\,x]^k\right)^n dx$$

- **Program code:**

```
Int[sin[c_.+d_.*x_]^m_*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  -Cos[c+d*x]*Sin[c+d*x]^((k-1)/2)*(a+b*Sin[c+d*x]^k)^(n+1)/(b*d*(n+2)) +
  Dist[1/(b*(n+2)),
    Int[Sin[c+d*x]^((k-1)/2)*(b*(n+1)-a*Sin[c+d*x]^k)*(a+b*Sin[c+d*x]^k)^n,x]] /;
FreeQ[{a,b,c,d},x] && OneQ[k^2] && ZeroQ[m-(5*k-1)/2] && RationalQ[n] && n>-1
```

Rules 1 – 6: $\int \left(\text{Sin}[c+d\,x]^j\right)^m \left(a+b\,\text{Sin}[c+d\,x]^k\right)^n \,dx$

- **Derivation: Recurrence 1 with** $A = 0$, $B = 0$, $C = 1$ **and** $m = m-2$

- **Rule 1a: If** $j^2 = k^2 = 1 \wedge a^2-b^2 \neq 0 \wedge j\,k\,m > 2 \wedge n < -1$, **then**

$$\int \left(\text{Sin}[c+d\,x]^j\right)^m \left(a+b\,\text{Sin}[c+d\,x]^k\right)^n \,dx \longrightarrow$$

$$-\frac{a^2\,\text{Cos}[c+d\,x]\,\left(\text{Sin}[c+d\,x]^j\right)^{m-2\,j\,k}\left(a+b\,\text{Sin}[c+d\,x]^k\right)^{n+1}}{b\,d\,(n+1)\,\left(a^2-b^2\right)}+\frac{1}{b\,(n+1)\,\left(a^2-b^2\right)}\cdot$$

$$\int \left(\text{Sin}[c+d\,x]^j\right)^{m-3\,j\,k}$$

$$\left(a^2\left(j\,k\,m+\frac{k-1}{2}-2\right)+a\,b\,(n+1)\,\text{Sin}[c+d\,x]^k-\left(b^2\,(n+1)+a^2\left(j\,k\,m+\frac{k-1}{2}-1\right)\right)\text{Sin}[c+d\,x]^{2\,k}\right)$$

$$\left(a+b\,\text{Sin}[c+d\,x]^k\right)^{n+1}\,dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  -a^2*Cos[c+d*x]*(Sin[c+d*x]^j)^(m-2*j*k)*(a+b*Sin[c+d*x]^k)^(n+1)/(b*d*(n+1)*(a^2-b^2)) +
  Dist[1/(b*(n+1)*(a^2-b^2)),
     Int[(sin[c+d*x]^j)^(m-3*j*k)*
        Sim[a^2*(j*k*m+(k-1)/2-2)+a*b*(n+1)*sin[c+d*x]^k-
          (b^2*(n+1)+a^2*(j*k*m+(k-1)/2-1))*sin[c+d*x]^(2*k),x]*
        (a+b*sin[c+d*x]^k)^(n+1),x]] /;
FreeQ[{a,b,c,d},x] && OneQ[j^2,k^2] && NonzeroQ[a^2-b^2] && RationalQ[m,n] && j*k*m>2 && n<-1
```

- **Derivation: Recurrence 1 with $A = 0$, $B = 1$, $C = 0$ and $m = m - 1$**
- **Derivation: Recurrence 6 with $A = 0$, $B = 0$, $C = 1$ and $m = m - 2$**
- **Rule 1b: If $j^2 = k^2 = 1 \land a^2 - b^2 \neq 0 \land 1 < j\,k\,m < 2 \land n < -1$, then**

$$\int \left(\text{Sin}[c+d\,x]^j\right)^m \left(a+b\,\text{Sin}[c+d\,x]^k\right)^n dx \longrightarrow$$

$$\frac{a\,\text{Cos}[c+d\,x]\,\left(\text{Sin}[c+d\,x]^j\right)^{m-j\,k}\left(a+b\,\text{Sin}[c+d\,x]^k\right)^{n+1}}{d\,(n+1)\,\left(a^2-b^2\right)} - \frac{1}{(n+1)\,\left(a^2-b^2\right)}\cdot$$

$$\int \left(\text{Sin}[c+d\,x]^j\right)^{m-2\,j\,k}\left(a\left(j\,k\,m+\frac{k-1}{2}-1\right)+b\,(n+1)\,\text{Sin}[c+d\,x]^k - a\left(j\,k\,m+n+\frac{k+1}{2}\right)\text{Sin}[c+d\,x]^{2\,k}\right)$$
$$\left(a+b\,\text{Sin}[c+d\,x]^k\right)^{n+1} dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  a*Cos[c+d*x]*(Sin[c+d*x]^j)^(m-j*k)*(a+b*Sin[c+d*x]^k)^(n+1)/(d*(n+1)*(a^2-b^2)) -
  Dist[1/((n+1)*(a^2-b^2)),
     Int[(sin[c+d*x]^j)^(m-2*j*k)*
       Sim[a*(j*k*m+(k-1)/2-1)+b*(n+1)*sin[c+d*x]^k-a*(j*k*m+n+(k+1)/2)*sin[c+d*x]^(2*k),x]*
       (a+b*sin[c+d*x]^k)^(n+1),x]] /;
FreeQ[{a,b,c,d},x] && OneQ[j^2,k^2] && NonzeroQ[a^2-b^2] && RationalQ[m,n] &&
  1<j*k*m<2 && n<-1
```

- **Derivation: Recurrence 1 with** $A = 1$, $B = 0$ **and** $C = 0$
- **Derivation: Recurrence 6 with** $A = 0$, $B = 1$, $C = 0$ **and** $m = m - 1$
- **Rule 1c: If** $j^2 = k^2 = 1 \wedge a^2 - b^2 \neq 0 \wedge 0 < j\,k\,m < 1 \wedge n < -1$, **then**

$$\int \left(\mathrm{Sin}[c+d\,x]^j\right)^m \left(a+b\,\mathrm{Sin}[c+d\,x]^k\right)^n \,dx \longrightarrow$$

$$-\frac{b\,\mathrm{Cos}[c+d\,x]\,\left(\mathrm{Sin}[c+d\,x]^j\right)^m \left(a+b\,\mathrm{Sin}[c+d\,x]^k\right)^{n+1}}{d\,(n+1)\,\left(a^2-b^2\right)}+\frac{1}{(n+1)\,\left(a^2-b^2\right)}\cdot$$

$$\int \left(\mathrm{Sin}[c+d\,x]^j\right)^{m-j\,k} \left(b\left(j\,k\,m+\frac{k-1}{2}\right)+a\,(n+1)\,\mathrm{Sin}[c+d\,x]^k-b\left(j\,k\,m+n+\frac{k+1}{2}+1\right)\mathrm{Sin}[c+d\,x]^{2\,k}\right)$$
$$\left(a+b\,\mathrm{Sin}[c+d\,x]^k\right)^{n+1}\,dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  -b*Cos[c+d*x]*(Sin[c+d*x]^j)^m*(a+b*Sin[c+d*x]^k)^(n+1)/(d*(n+1)*(a^2-b^2)) +
  Dist[1/((n+1)*(a^2-b^2)),
    Int[(sin[c+d*x]^j)^(m-j*k)*
      Sim[b*(j*k*m+(k-1)/2)+a*(n+1)*sin[c+d*x]^k-b*(j*k*m+n+(k+1)/2+1)*sin[c+d*x]^(2*k),x]*
      (a+b*sin[c+d*x]^k)^(n+1),x]] /;
FreeQ[{a,b,c,d},x] && OneQ[j^2,k^2] && NonzeroQ[a^2-b^2] && RationalQ[m,n] &&
  0<j*k*m<1 && n<-1
```

- **Derivation: Recurrence 2 with $A = 0$, $B = 0$, $C = 1$ and $m = m - 2$**
- **Rule 2a: If $j^2 = k^2 = 1 \land a^2 - b^2 \neq 0 \land j\,k\,m > 2 \land -1 \le n < 0$, then**

$$\int \left(\mathrm{Sin}[c+d\,x]^j\right)^m \left(a+b\,\mathrm{Sin}[c+d\,x]^k\right)^n dx \longrightarrow$$

$$-\frac{\mathrm{Cos}[c+d\,x]\left(\mathrm{Sin}[c+d\,x]^j\right)^{m-2\,j\,k}\left(a+b\,\mathrm{Sin}[c+d\,x]^k\right)^{n+1}}{b\,d\left(j\,k\,m+n+\frac{k-1}{2}\right)}+\frac{1}{b\left(j\,k\,m+n+\frac{k-1}{2}\right)}\cdot$$

$$\int \left(\mathrm{Sin}[c+d\,x]^j\right)^{m-3\,j\,k}$$

$$\left(a\left(j\,k\,m+\frac{k-1}{2}-2\right)+b\left(j\,k\,m+n+\frac{k-1}{2}-1\right)\mathrm{Sin}[c+d\,x]^k-a\left(j\,k\,m+\frac{k-1}{2}-1\right)\mathrm{Sin}[c+d\,x]^{2\,k}\right)$$

$$\left(a+b\,\mathrm{Sin}[c+d\,x]^k\right)^n dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_*(a_.+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  -Cos[c+d*x]*(Sin[c+d*x]^j)^(m-2*j*k)*(a+b*Sin[c+d*x]^k)^(n+1)/(b*d*(j*k*m+n+(k-1)/2)) +
  Dist[1/(b*(j*k*m+n+(k-1)/2)),
    Int[(sin[c+d*x]^j)^(m-3*j*k)*
      Sim[a*(j*k*m+(k-1)/2-2)+b*(j*k*m+n+(k-1)/2-1)*sin[c+d*x]^k-
        a*(j*k*m+(k-1)/2-1)*sin[c+d*x]^(2*k),x]*
      (a+b*sin[c+d*x]^k)^n,x]] /;
FreeQ[{a,b,c,d},x] && OneQ[j^2,k^2] && NonzeroQ[a^2-b^2] && RationalQ[m,n] && j*k*m>2 && -1≤n<0
```

- **Derivation: Recurrence 2 with** $A = 0$, $B = a$, $C = b$, $m = m - 1$ **and** $n = n - 1$
- **Derivation: Recurrence 3 with** $A = 0$, $B = 0$, $C = 1$ **and** $m = m - 2$
- **Rule 2b: If** $j^2 = k^2 = 1 \wedge a^2 - b^2 \neq 0 \wedge j\,k\,m > 1 \wedge j\,k\,m \neq 2 \wedge 0 < n < 1$, **then**

$$\int \left(\text{Sin}[c+d\,x]^j\right)^m \left(a+b\,\text{Sin}[c+d\,x]^k\right)^n dx \longrightarrow$$

$$-\frac{\text{Cos}[c+d\,x]\,\left(\text{Sin}[c+d\,x]^j\right)^{m-j\,k}\,\left(a+b\,\text{Sin}[c+d\,x]^k\right)^n}{d\left(j\,k\,m+n+\frac{k-1}{2}\right)}+\frac{1}{j\,k\,m+n+\frac{k-1}{2}}\cdot$$

$$\int \left(\text{Sin}[c+d\,x]^j\right)^{m-2\,j\,k}$$

$$\left(a\left(j\,k\,m+\frac{k-1}{2}-1\right)+b\left(j\,k\,m+n+\frac{k-1}{2}-1\right)\text{Sin}[c+d\,x]^k+a\,n\,\text{Sin}[c+d\,x]^{2\,k}\right)\left(a+b\,\text{Sin}[c+d\,x]^k\right)^{n-1}dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_*(a_.+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  -Cos[c+d*x]*(Sin[c+d*x]^j)^(m-j*k)*(a+b*Sin[c+d*x]^k)^n/(d*(j*k*m+n+(k-1)/2)) +
  Dist[1/(j*k*m+n+(k-1)/2),
     Int[(sin[c+d*x]^j)^(m-2*j*k)*
        Sim[a*(j*k*m+(k-1)/2-1)+b*(j*k*m+n+(k-1)/2-1)*sin[c+d*x]^k+a*n*sin[c+d*x]^(2*k),x]*
        (a+b*sin[c+d*x]^k)^(n-1),x]] /;
FreeQ[{a,b,c,d},x] && OneQ[j^2,k^2] && NonzeroQ[a^2-b^2] && RationalQ[m,n] &&
  j*k*m>1 && j*k*m≠2 && 0<n<1
```

- **Derivation: Recurrence 3 with** $A = a^2$, $B = 2\,a\,b$, $C = b^2$ **and** $n = n - 2$

- **Rule 3a: If** $j^2 = k^2 = 1 \wedge a^2 - b^2 \neq 0 \wedge j\,k\,m \geq -1 \wedge j\,k\,m \neq 1 \wedge j\,k\,m \neq 2 \wedge n > 2$**, then**

$$\int \left(\text{Sin}[c+d\,x]^j\right)^m \left(a+b\,\text{Sin}[c+d\,x]^k\right)^n \,dx \longrightarrow$$

$$-\frac{b^2\,\text{Cos}[c+d\,x]\,\left(\text{Sin}[c+d\,x]^j\right)^{m+j\,k}\left(a+b\,\text{Sin}[c+d\,x]^k\right)^{n-2}}{d\left(j\,k\,m+n+\frac{k-1}{2}\right)}+\frac{1}{j\,k\,m+n+\frac{k-1}{2}}\cdot$$

$$\int\left(\text{Sin}[c+d\,x]^j\right)^m$$

$$\left(a\left(a^2\,(n-1)+\left(a^2+b^2\right)\left(j\,k\,m+\frac{k+1}{2}\right)\right)+b\left(-b^2+\left(3\,a^2+b^2\right)\left(j\,k\,m+n+\frac{k-1}{2}\right)\right)\text{Sin}[c+d\,x]^k+\right.$$

$$\left.a\,b^2\,(2\,j\,k\,m+3\,n+k-3)\,\text{Sin}[c+d\,x]^{2\,k}\right)\left(a+b\,\text{Sin}[c+d\,x]^k\right)^{n-3}dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(a_.+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  -b^2*Cos[c+d*x]*(Sin[c+d*x]^j)^(m+j*k)*(a+b*Sin[c+d*x]^k)^(n-2)/(d*(j*k*m+n+(k-1)/2)) +
  Dist[1/(j*k*m+n+(k-1)/2),
    Int[(sin[c+d*x]^j)^m*
      Sim[a*(a^2*(n-1)+(a^2+b^2)*(j*k*m+(k+1)/2))+b*(-b^2+(3*a^2+b^2)*(j*k*m+n+(k-1)/2))*sin[c+d*x]^k+
        a*b^2*(2*j*k*m+3*n+k-3)*sin[c+d*x]^(2*k),x]*
      (a+b*sin[c+d*x]^k)^(n-3),x]] /;
FreeQ[{a,b,c,d},x] && OneQ[j^2,k^2] && NonzeroQ[a^2-b^2] && RationalQ[m,n] &&
  j*k*m≥-1 && j*k*m≠1 && j*k*m≠2 && n>2
```

- **Derivation: Recurrence 3 with $A = 0$, $B = a$, $C = b$, $m = m - 1$ and $n = n - 1$**

- **Rule 3b: If $j^2 = k^2 = 1 \land a^2 - b^2 \neq 0 \land j\,k\,m > 0 \land j\,k\,m \neq 1 \land j\,k\,m \neq 2 \land 1 < n < 2$, then**

$$\int \left(\text{Sin}[c+d\,x]^j\right)^m \left(a+b\,\text{Sin}[c+d\,x]^k\right)^n dx \longrightarrow$$

$$-\frac{b\,\text{Cos}[c+d\,x]\left(\text{Sin}[c+d\,x]^j\right)^m\left(a+b\,\text{Sin}[c+d\,x]^k\right)^{n-1}}{d\left(j\,k\,m+n+\frac{k-1}{2}\right)}+\frac{1}{j\,k\,m+n+\frac{k-1}{2}}\cdot$$

$$\int\left(\text{Sin}[c+d\,x]^j\right)^{m-j\,k}\left(a\,b\left(j\,k\,m+\frac{k-1}{2}\right)+\left(\left(a^2+b^2\right)\left(j\,k\,m+n+\frac{k-1}{2}\right)-b^2\right)\text{Sin}[c+d\,x]^k+\right.$$
$$\left.a\,b\left(j\,k\,m+2\,n+\frac{k-1}{2}-1\right)\text{Sin}[c+d\,x]^{2\,k}\right)\left(a+b\,\text{Sin}[c+d\,x]^k\right)^{n-2}dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(a_.+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  -b*Cos[c+d*x]*(Sin[c+d*x]^j)^m*(a+b*Sin[c+d*x]^k)^(n-1)/(d*(j*k*m+n+(k-1)/2)) +
  Dist[1/(j*k*m+n+(k-1)/2),
    Int[(sin[c+d*x]^j)^(m-j*k)*
      Sim[a*b*(j*k*m+(k-1)/2)+((a^2+b^2)*(j*k*m+n+(k-1)/2)-b^2)*sin[c+d*x]^k+
        a*b*(j*k*m+2*n+(k-1)/2-1)*sin[c+d*x]^(2*k),x]*
      (a+b*sin[c+d*x]^k)^(n-2),x]] /;
FreeQ[{a,b,c,d},x] && OneQ[j^2,k^2] && NonzeroQ[a^2-b^2] && RationalQ[m,n] &&
  j*k*m>0 && j*k*m≠1 && j*k*m≠2 && 1<n<2
```

- **Derivation: Recurrence 4 with** $A = a^2$, $B = 2\,a\,b$, $C = b^2$ **and** $n = n-2$

- **Rule 4a: If** $j^2 = k^2 = 1 \bigwedge a^2 - b^2 \neq 0 \bigwedge j\,k\,m < -1 \bigwedge n > 2$, **then**

$$\int \left(\mathrm{Sin}[c+d\,x]^j\right)^m \left(a+b\,\mathrm{Sin}[c+d\,x]^k\right)^n dx \longrightarrow$$

$$\frac{a^2\,\mathrm{Cos}[c+d\,x]\,\left(\mathrm{Sin}[c+d\,x]^j\right)^{m+j\,k}\left(a+b\,\mathrm{Sin}[c+d\,x]^k\right)^{n-2}}{d\left(j\,k\,m+\frac{k+1}{2}\right)}+\frac{1}{j\,k\,m+\frac{k+1}{2}}\cdot$$

$$\int \left(\mathrm{Sin}[c+d\,x]^j\right)^{m+j\,k}\left(a^2\,b\,(2\,j\,k\,m-n+k+3)+a\left(a^2+\left(a^2+3\,b^2\right)\left(j\,k\,m+\frac{k+1}{2}\right)\right)\mathrm{Sin}[c+d\,x]^k+\right.$$
$$\left.b\left(a^2\,(n-1)+\left(a^2+b^2\right)\left(j\,k\,m+\frac{k+1}{2}\right)\right)\mathrm{Sin}[c+d\,x]^{2\,k}\right)\left(a+b\,\mathrm{Sin}[c+d\,x]^k\right)^{n-3} dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  a^2*Cos[c+d*x]*(Sin[c+d*x]^j)^(m+j*k)*(a+b*Sin[c+d*x]^k)^(n-2)/(d*(j*k*m+(k+1)/2)) +
  Dist[1/(j*k*m+(k+1)/2),
    Int[(sin[c+d*x]^j)^(m+j*k)*
      Sim[a^2*b*(2*j*k*m-n+k+3)+a*(a^2+(a^2+3*b^2)*(j*k*m+(k+1)/2))*sin[c+d*x]^k+
        b*(a^2*(n-1)+(a^2+b^2)*(j*k*m+(k+1)/2))*sin[c+d*x]^(2*k),x]*
      (a+b*sin[c+d*x]^k)^(n-3),x]] /;
FreeQ[{a,b,c,d},x] && OneQ[j^2,k^2] && NonzeroQ[a^2-b^2] && RationalQ[m,n] &&
  j*k*m<-1 && n>2
```

- **Derivation: Recurrence 4 with** $A = a$, $B = b$, $C = 0$ **and** $n = n - 1$

- **Rule 4b: If** $j^2 = k^2 = 1 \land a^2 - b^2 \neq 0 \land j\,k\,m < -1 \land 1 < n < 2$, **then**

$$\int \left(\text{Sin}[c+d\,x]^j\right)^m \left(a+b\,\text{Sin}[c+d\,x]^k\right)^n dx \longrightarrow$$

$$\frac{a\,\text{Cos}[c+d\,x] \left(\text{Sin}[c+d\,x]^j\right)^{m+j\,k} \left(a+b\,\text{Sin}[c+d\,x]^k\right)^{n-1}}{d\left(j\,k\,m+\frac{k+1}{2}\right)} + \frac{1}{j\,k\,m+\frac{k+1}{2}}\,\cdot$$

$$\int \left(\text{Sin}[c+d\,x]^j\right)^{m+j\,k} \left(a\,b\left(j\,k\,m-n+\frac{k+1}{2}+1\right) + \left(a^2+\left(a^2+b^2\right)\left(j\,k\,m+\frac{k+1}{2}\right)\right)\text{Sin}[c+d\,x]^k +\right.$$

$$\left. a\,b\left(j\,k\,m+n+\frac{k+1}{2}\right)\text{Sin}[c+d\,x]^{2\,k}\right)\left(a+b\,\text{Sin}[c+d\,x]^k\right)^{n-2} dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  a*Cos[c+d*x]*(Sin[c+d*x]^j)^(m+j*k)*(a+b*Sin[c+d*x]^k)^(n-1)/(d*(j*k*m+(k+1)/2)) +
  Dist[1/(j*k*m+(k+1)/2),
     Int[(sin[c+d*x]^j)^(m+j*k)*
       Sim[a*b*(j*k*m-n+(k+1)/2+1)+(a^2+(a^2+b^2)*(j*k*m+(k+1)/2))*sin[c+d*x]^k+
         a*b*(j*k*m+n+(k+1)/2)*sin[c+d*x]^(2*k),x]*
       (a+b*sin[c+d*x]^k)^(n-2),x]] /;
FreeQ[{a,b,c,d},x] && OneQ[j^2,k^2] && NonzeroQ[a^2-b^2] && RationalQ[m,n] &&
  j*k*m<-1 && 1<n<2
```

- **Derivation: Recurrence 4 with** $A = 1$, $B = 0$ **and** $C = 0$

- **Derivation: Recurrence 5 with** $A = a$, $B = b$, $C = 0$ **and** $n = n - 1$

- **Rule 5a: If** $j^2 = k^2 = 1 \bigwedge a^2 - b^2 \neq 0 \bigwedge j\,k\,m + \frac{k+1}{2} \neq 0 \bigwedge j\,k\,m \leq -1 \bigwedge 0 < n < 1$, **then**

$$\int \left(\mathrm{Sin}[c + d\,x]^j\right)^m \left(a + b\,\mathrm{Sin}[c + d\,x]^k\right)^n \,dx \longrightarrow$$

$$\frac{\mathrm{Cos}[c + d\,x] \left(\mathrm{Sin}[c + d\,x]^j\right)^{m+j\,k} \left(a + b\,\mathrm{Sin}[c + d\,x]^k\right)^n}{d \left(j\,k\,m + \frac{k+1}{2}\right)} + \frac{1}{j\,k\,m + \frac{k+1}{2}} \cdot$$

$$\int \left(\mathrm{Sin}[c + d\,x]^j\right)^{m+j\,k} \left(-b\,n + a \left(j\,k\,m + \frac{k+1}{2} + 1\right) \mathrm{Sin}[c + d\,x]^k + b \left(j\,k\,m + n + \frac{k+1}{2} + 1\right) \mathrm{Sin}[c + d\,x]^{2\,k}\right)$$
$$\left(a + b\,\mathrm{Sin}[c + d\,x]^k\right)^{n-1} \,dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(a_.+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  Cos[c+d*x]*(Sin[c+d*x]^j)^(m+j*k)*(a+b*Sin[c+d*x]^k)^n/(d*(j*k*m+(k+1)/2)) +
  Dist[1/(j*k*m+(k+1)/2),
    Int[(sin[c+d*x]^j)^(m+j*k)*
      Sim[-b*n+a*(j*k*m+(k+1)/2+1)*sin[c+d*x]^k+b*(j*k*m+n+(k+1)/2+1)*sin[c+d*x]^(2*k),x]*
      (a+b*sin[c+d*x]^k)^(n-1),x]] /;
FreeQ[{a,b,c,d},x] && OneQ[j^2,k^2] && NonzeroQ[a^2-b^2] && RationalQ[m,n] &&
  j*k*m+(k+1)/2≠0 && j*k*m≤-1 && 0<n<1
```

- **Derivation: Recurrence 5 with $A = 1$, $B = 0$ and $C = 0$**

- **Rule 5b: If $j^2 = k^2 = 1 \bigwedge a^2 - b^2 \neq 0 \bigwedge j\,k\,m + \frac{k+1}{2} \neq 0 \bigwedge j\,k\,m \leq -1 \bigwedge -1 \leq n < 0$, then**

$$\int \left(\text{Sin}[c + d\,x]^j\right)^m \left(a + b\,\text{Sin}[c + d\,x]^k\right)^n dx \longrightarrow$$

$$\frac{\text{Cos}[c + d\,x] \left(\text{Sin}[c + d\,x]^j\right)^{m+j\,k} \left(a + b\,\text{Sin}[c + d\,x]^k\right)^{n+1}}{a\,d \left(j\,k\,m + \frac{k+1}{2}\right)} + \frac{1}{a \left(j\,k\,m + \frac{k+1}{2}\right)} \cdot$$

$$\int \left(\text{Sin}[c + d\,x]^j\right)^{m+j\,k}$$

$$\left(-b \left(j\,k\,m + n + \frac{k+1}{2} + 1\right) + a \left(j\,k\,m + \frac{k+1}{2} + 1\right) \text{Sin}[c + d\,x]^k + b \left(j\,k\,m + n + \frac{k+1}{2} + 2\right) \text{Sin}[c + d\,x]^{2\,k}\right)$$

$$\left(a + b\,\text{Sin}[c + d\,x]^k\right)^n dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  Cos[c+d*x]*(Sin[c+d*x]^j)^(m+j*k)*(a+b*Sin[c+d*x]^k)^(n+1)/(a*d*(j*k*m+(k+1)/2)) +
  Dist[1/(a*(j*k*m+(k+1)/2)),
    Int[(sin[c+d*x]^j)^(m+j*k)*
      Sim[-b*(j*k*m+n+(k+1)/2+1)+a*(j*k*m+(k+1)/2+1)*sin[c+d*x]^k+
        b*(j*k*m+n+(k+1)/2+2)*sin[c+d*x]^(2*k),x]*
      (a+b*sin[c+d*x]^k)^n,x]] /;
FreeQ[{a,b,c,d},x] && OneQ[j^2,k^2] && NonzeroQ[a^2-b^2] && RationalQ[m,n] &&
  j*k*m+(k+1)/2≠0 && j*k*m≤-1 && -1≤n<0
```

- **Derivation: Recurrence 6 with $A = 1$, $B = 0$ and $C = 0$**

- **Rule 6: If $j^2 = k^2 = 1 \land a^2 - b^2 \neq 0 \land j\,k\,m < 0 \land n < -1$, then**

$$\int \left(\text{Sin}[c + d\,x]^j\right)^m \left(a + b\,\text{Sin}[c + d\,x]^k\right)^n \, dx \longrightarrow$$

$$\frac{b^2 \,\text{Cos}[c + d\,x] \left(\text{Sin}[c + d\,x]^j\right)^{m+j\,k} \left(a + b\,\text{Sin}[c + d\,x]^k\right)^{n+1}}{a\,d\,(n + 1)\left(a^2 - b^2\right)} + \frac{1}{a\,(n + 1)\left(a^2 - b^2\right)} \cdot$$

$$\int \left(\text{Sin}[c + d\,x]^j\right)^m \left(\left(a^2 - b^2\right)(n + 1) - b^2\left(j\,k\,m + \frac{k + 1}{2}\right) -\right.$$

$$\left. a\,b\,(n + 1)\,\text{Sin}[c + d\,x]^k + b^2\left(j\,k\,m + n + \frac{k + 1}{2} + 2\right)\text{Sin}[c + d\,x]^{2\,k}\right)\left(a + b\,\text{Sin}[c + d\,x]^k\right)^{n+1} dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  b^2*Cos[c+d*x]*(Sin[c+d*x]^j)^(m+j*k)*(a+b*Sin[c+d*x]^k)^(n+1)/(a*d*(n+1)*(a^2-b^2)) +
  Dist[1/(a*(n+1)*(a^2-b^2)),
    Int[(sin[c+d*x]^j)^m*
      Sim[(a^2-b^2)*(n+1)-b^2*(j*k*m+(k+1)/2)-a*b*(n+1)*sin[c+d*x]^k+
        b^2*(j*k*m+n+(k+1)/2+2)*sin[c+d*x]^(2*k),x]*
      (a+b*sin[c+d*x]^k)^(n+1),x]] /;
FreeQ[{a,b,c,d},x] && OneQ[j^2,k^2] && NonzeroQ[a^2-b^2] && RationalQ[m,n] &&
  j*k*m<0 && n<-1
```