$$\int \text{Sech}[a + b\,x]^n\, dx$$

- **Reference: G&R 2.423.9, CRC 558, A&S 4.5.81**
- **Derivation: Integration by substitution**
- **Basis:** $\text{Sech}[z] = \frac{\text{Sinh}'[z]}{1+\text{Sinh}[z]^2}$
- **Rule:**

$$\int \text{Sech}[a + b\,x]\, dx \longrightarrow \frac{\text{ArcTan}[\text{Sinh}[a + b\,x]]}{b}$$

- **Program code:**

```
Int[Sech[a_.+b_.*x_],x_Symbol] :=
(* -ArcCot[Sinh[a+b*x]]/b *)
  ArcTan[Sinh[a+b*x]]/b /;
FreeQ[{a,b},x]
```

- **Reference: G&R 2.423.1', CRC 559', A&S 4.5.80'**

```
Int[Csch[a_.+b_.*x_],x_Symbol] :=
(* -ArcTanh[Cosh[a+b*x]]/b *)
  -ArcCoth[Cosh[a+b*x]]/b /;
FreeQ[{a,b},x]
```

- **Reference: G&R 2.423.10, CRC 571**
- **Derivation: Primitive rule**
- **Basis:** $\text{Tanh}'[z] = \text{Sech}[z]^2$
- **Rule:**

$$\int \text{Sech}[a + b\,x]^2\, dx \longrightarrow \frac{\text{Tanh}[a + b\,x]}{b}$$

- **Program code:**

```
Int[Sech[a_.+b_.*x_]^2,x_Symbol] :=
  Tanh[a+b*x]/b /;
FreeQ[{a,b},x]
```

- **Reference: G&R 2.423.2, CRC 575**

```
Int[Csch[a_.+b_.*x_]^2,x_Symbol] :=
  -Coth[a+b*x]/b /;
FreeQ[{a,b},x]
```

- **Derivation: Integration by substitution**
- **Basis: If $\frac{n}{2} \in \mathbb{Z}$, then $\text{Sech}[z]^n = \left(1 - \text{Tanh}[z]^2\right)^{\frac{n-2}{2}} \text{Tanh}'[z]$**
- **Note: This rule is used for even n since it requires fewer steps and results in a simpler antiderivative than the recursive rule.**
- **Rule: If $\frac{n}{2} \in \mathbb{Z} \bigwedge n > 1$, then**

$$\int \text{Sech}[a + b\,x]^n \, dx \longrightarrow \frac{1}{b} \,\text{Subst}\left[\int \left(1 - x^2\right)^{\frac{n-2}{2}} dx,\, x,\, \text{Cosh}[a + b\,x]\right]$$

- **Program code:**

```
Int[Sech[a_.+b_.*x_]^n_,x_Symbol] :=
  Dist[1/b,Subst[Int[Regularize[(1-x^2)^((n-2)/2),x],x],x,Tanh[a+b*x]]] /;
FreeQ[{a,b},x] && EvenQ[n] && n>1
```

```
Int[Csch[a_.+b_.*x_]^n_,x_Symbol] :=
  Dist[-1/b,Subst[Int[Regularize[(-1+x^2)^((n-2)/2),x],x],x,Coth[a+b*x]]] /;
FreeQ[{a,b},x] && EvenQ[n] && n>1
```

- **Reference: G&R 2.411.6, CRC 568b**
- **Derivation: Integration by parts with a double-back flip**
- **Rule: If $\frac{n}{2} \in \mathbb{Z} \bigwedge n > 1$, then**

$$\int \text{Sech}[a + b\,x]^n \, dx \longrightarrow \frac{\text{Sinh}[a + b\,x]\, \text{Sech}[a + b\,x]^{n-1}}{b\,(n-1)} + \frac{n-2}{n-1} \int \text{Sech}[a + b\,x]^{n-2} \, dx$$

- **Program code:**

```
Int[Sech[a_.+b_.*x_]^n_,x_Symbol] :=
  Sinh[a+b*x]*Sech[a+b*x]^(n-1)/(b*(n-1)) +
  Dist[(n-2)/(n-1),Int[Sech[a+b*x]^(n-2),x]] /;
FreeQ[{a,b},x] && Not[EvenQ[n]] && RationalQ[n] && n>1
```

- **Reference: G&R 2.411.5, CRC 568a**

```
Int[Csch[a_.+b_.*x_]^n_,x_Symbol] :=
  -Cosh[a+b*x]*Csch[a+b*x]^(n-1)/(b*(n-1)) -
  Dist[(n-2)/(n-1),Int[Csch[a+b*x]^(n-2),x]] /;
FreeQ[{a,b},x] && Not[EvenQ[n]] && RationalQ[n] && n>1
```

$$\int (c\,\mathrm{Sech}[a+b\,x])^n\,dx$$

- **Derivation: Piecewise constant extraction**
- **Basis:** $\partial_z\left((c/f[z])^n\,f[z]^n\right)=0$
- **Note: The special case rules for** $c=1$ **and** $n=-\frac{1}{2}$ **are required due to an idem potent problem in Mathematica 6 & 7.**
- **Rule: If** $-1<n<1$**, then**

$$\int (c\,\mathrm{Sech}[a+b\,x])^n\,dx \;\longrightarrow\; (c\,\mathrm{Sech}[a+b\,x])^n\,\mathrm{Cosh}[a+b\,x]^n\int \frac{1}{\mathrm{Cosh}[a+b\,x]^n}\,dx$$

- **Program code:**

```
Int[1/Sqrt[Sech[a_.+b_.*x_]],x_Symbol] :=
  Sqrt[Cosh[a+b*x]]*Sqrt[Sech[a+b*x]]*Int[Sqrt[Cosh[a+b*x]],x] /;
FreeQ[{a,b},x]
```

```
Int[(c_.*Sech[a_.+b_.*x_])^n_,x_Symbol] :=
  (c*Sech[a+b*x])^n*Cosh[a+b*x]^n*Int[1/Cosh[a+b*x]^n,x] /;
FreeQ[{a,b,c},x] && RationalQ[n] && -1<n<1
```

```
Int[1/Sqrt[Csch[a_.+b_.*x_]],x_Symbol] :=
  Sqrt[Csch[a+b*x]]*Sqrt[Sinh[a+b*x]]*Int[Sqrt[Sinh[a+b*x]],x] /;
FreeQ[{a,b},x]
```

```
Int[(c_.*Csch[a_.+b_.*x_])^n_,x_Symbol] :=
  (c*Csch[a+b*x])^n*Sinh[a+b*x]^n*Int[1/Sinh[a+b*x]^n,x] /;
FreeQ[{a,b,c},x] && RationalQ[n] && -1<n<1
```

- **Reference: G&R 2.411.6, CRC 568b**
- **Derivation: Integration by parts with a double-back flip**
- **Rule: If** $n>1$**, then**

$$\int (c\,\mathrm{Sech}[a+b\,x])^n\,dx \;\longrightarrow\; \frac{c\,\mathrm{Sinh}[a+b\,x]\,(c\,\mathrm{Sech}[a+b\,x])^{n-1}}{b\,(n-1)}+\frac{c^2\,(n-2)}{n-1}\int (c\,\mathrm{Sech}[a+b\,x])^{n-2}\,dx$$

- **Program code:**

```
Int[(c_.*Sech[a_.+b_.*x_])^n_,x_Symbol] :=
  c*Sinh[a+b*x]*(c*Sech[a+b*x])^(n-1)/(b*(n-1)) +
  Dist[c^2*(n-2)/(n-1),Int[(c*Sech[a+b*x])^(n-2),x]] /;
FreeQ[{a,b,c},x] && FractionQ[n] && n>1
```

- **Reference: G&R 2.411.5, CRC 568a**

```
Int[(c_.*Csch[a_.+b_.*x_])^n_,x_Symbol] :=
  -c*Cosh[a+b*x]*(c*Csch[a+b*x])^(n-1)/(b*(n-1)) -
  Dist[c^2*(n-2)/(n-1),Int[(c*Csch[a+b*x])^(n-2),x]] /;
FreeQ[{a,b,c},x] && FractionQ[n] && n>1
```

- **Reference: G&R 2.411.1, CRC 567a**
- **Derivation: Integration by parts with a double-back flip**
- **Rule: If $n < -1$, then**

$$\int (c\,\mathrm{Sech}[a+b\,x])^n\,dx \longrightarrow -\frac{\mathrm{Sinh}[a+b\,x]\,(c\,\mathrm{Sech}[a+b\,x])^{n+1}}{b\,c\,n}+\frac{(n+1)}{c^2\,n}\int (c\,\mathrm{Sech}[a+b\,x])^{n+2}\,dx$$

- **Program code:**

```
Int[(c_.*Sech[a_.+b_.*x_])^n_,x_Symbol] :=
  -Sinh[a+b*x]*(c*Sech[a+b*x])^(n+1)/(b*c*n) +
  Dist[(n+1)/(c^2*n),Int[(c*Sech[a+b*x])^(n+2),x]] /;
FreeQ[{a,b,c},x] && FractionQ[n] && n<-1
```

- **Reference: G&R 2.411.2, CRC 567b**

```
Int[(c_.*Csch[a_.+b_.*x_])^n_,x_Symbol] :=
  -Cosh[a+b*x]*(c*Csch[a+b*x])^(n+1)/(b*c*n) -
  Dist[(n+1)/(c^2*n),Int[(c*Csch[a+b*x])^(n+2),x]] /;
FreeQ[{a,b,c},x] && FractionQ[n] && n<-1
```

$$\int (a + b\,\mathrm{Sech}[c + d\,x])^{n/2}\,dx$$

- **Rule: If** $a^2 - b^2 = 0$**, then**

$$\int \sqrt{a + b\,\mathrm{Sech}[c + d\,x]}\,dx \longrightarrow \frac{2\,a\,\mathrm{ArcTan}\left[\sqrt{-1 + \frac{a\,\mathrm{Sech}[c+d\,x]}{b}}\right]\mathrm{Tanh}[c + d\,x]}{d\sqrt{-1 + \frac{a\,\mathrm{Sech}[c+d\,x]}{b}}\,\sqrt{a + b\,\mathrm{Sech}[c + d\,x]}}$$

- **Program code:**

```
Int[Sqrt[a_+b_.*Sech[c_.+d_.*x_]],x_Symbol] :=
  2*a*ArcTan[Sqrt[-1+a/b*Sech[c+d*x]]]*Tanh[c+d*x]/
        (d*Sqrt[-1+a/b*Sech[c+d*x]]*Sqrt[a+b*Sech[c+d*x]]) /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2-b^2]
```

```
Int[Sqrt[a_+b_.*Csch[c_.+d_.*x_]],x_Symbol] :=
  2*a*ArcTan[Sqrt[-1-a/b*Csch[c+d*x]]]*Coth[c+d*x]/
        (d*Sqrt[-1-a/b*Csch[c+d*x]]*Sqrt[a+b*Csch[c+d*x]]) /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2+b^2]
```

- **Note: Is there a simpler antiderivative?**
- **Rule: If** $a^2 - b^2 = 0$**, then**

$$\int \frac{1}{\sqrt{a + b\,\mathrm{Sech}[c + d\,x]}}\,dx \longrightarrow$$
$$-\frac{\mathrm{Coth}[c + d\,x]\,\sqrt{-a + b\,\mathrm{Sech}[c + d\,x]}\,\sqrt{a + b\,\mathrm{Sech}[c + d\,x]}}{a^{3/2}\,d}$$
$$\left(\sqrt{2}\,\mathrm{ArcTan}\left[\frac{\sqrt{2\,a}}{\sqrt{-a + b\,\mathrm{Sech}[c + d\,x]}}\right] + 2\,\mathrm{ArcTan}\left[\frac{\sqrt{-a + b\,\mathrm{Sech}[c + d\,x]}}{\sqrt{a}}\right]\right)$$

- **Program code:**

```
Int[1/Sqrt[a_+b_.*Sech[c_.+d_.*x_]],x_Symbol] :=
  -Coth[c+d*x]*Sqrt[-a+b*Sech[c+d*x]]*Sqrt[a+b*Sech[c+d*x]]/(a^(3/2)*d)*
    (Sqrt[2]*ArcTan[Sqrt[2*a]/Sqrt[-a+b*Sech[c+d*x]]]+2*ArcTan[Sqrt[-a+b*Sech[c+d*x]]/Sqrt[a]]) /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2-b^2]
```

```
Int[1/Sqrt[a_+b_.*Csch[c_.+d_.*x_]],x_Symbol] :=
  -Sqrt[-a+b*Csch[c+d*x]]*Sqrt[a+b*Csch[c+d*x]]*Tanh[c+d*x]/a^(3/2)*
    (Sqrt[2]*ArcTan[Sqrt[2*a]/Sqrt[-a+b*Csch[c+d*x]]]+2*ArcTan[Sqrt[-a+b*Csch[c+d*x]]/Sqrt[a]]) /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2+b^2]
```

$$\int x^m \operatorname{Sech}[a+b\,x]^n \, dx$$

- **Derivation: Integration by parts**

- **Rule: If $m \in \mathbb{Z} \land m > 0$, then**

$$\int x^m \operatorname{Sech}[a+b\,x] \, dx \longrightarrow \frac{2\,x^m \operatorname{ArcTan}\left[e^{a+b\,x}\right]}{b} - \frac{2\,m}{b} \int x^{m-1} \operatorname{ArcTan}\left[e^{a+b\,x}\right] dx$$

- **Program code:**

```
Int[x_^m_.*Sech[a_.+b_.*x_],x_Symbol] :=
  2*x^m*ArcTan[E^(a+b*x)]/b -
  Dist[2*m/b,Int[x^(m-1)*ArcTan[E^(a+b*x)],x]] /;
FreeQ[{a,b},x] && IntegerQ[m] && m>0
```

```
Int[x_^m_.*Csch[a_.+b_.*x_],x_Symbol] :=
  -2*x^m*ArcTanh[E^(a+b x)]/b +
  Dist[2*m/b,Int[x^(m-1)*ArcTanh[E^(a+b x)],x]] /;
FreeQ[{a,b},x] && IntegerQ[m] && m>0
```

- **Reference: CRC 430h**

- **Rule: If $m > 0$, then**

$$\int x^m \operatorname{Sech}[a+b\,x]^2 \, dx \longrightarrow \frac{x^m \operatorname{Tanh}[a+b\,x]}{b} - \frac{m}{b} \int x^{m-1} \operatorname{Tanh}[a+b\,x] \, dx$$

- **Program code:**

```
Int[x_^m_.*Sech[a_.+b_.*x_]^2,x_Symbol] :=
  x^m*Tanh[a+b*x]/b -
  Dist[m/b,Int[x^(m-1)*Tanh[a+b*x],x]] /;
FreeQ[{a,b},x] && RationalQ[m] && m>0
```

- **Reference: CRC 428h**

```
Int[x_^m_.*Csch[a_.+b_.*x_]^2,x_Symbol] :=
  -x^m*Coth[a+b*x]/b +
  Dist[m/b,Int[x^(m-1)*Coth[a+b*x],x]] /;
FreeQ[{a,b},x] && RationalQ[m] && m>0
```

- **Reference: G&R 2.643.2h, CRC 431h**
- **Rule: If $n > 1 \land n \neq 2$, then**

$$\int x\,\text{Sech}[a+b\,x]^n\,dx \longrightarrow \frac{x\,\text{Tanh}[a+b\,x]\,\text{Sech}[a+b\,x]^{n-2}}{b\,(n-1)} + \frac{\text{Sech}[a+b\,x]^{n-2}}{b^2\,(n-1)\,(n-2)} + \frac{n-2}{n-1}\int x\,\text{Sech}[a+b\,x]^{n-2}\,dx$$

- **Program code:**

```
Int[x_*Sech[a_.+b_.*x_]^n_,x_Symbol] :=
  x*Tanh[a+b*x]*Sech[a+b*x]^(n-2)/(b*(n-1)) +
  Sech[a+b*x]^(n-2)/(b^2*(n-1)*(n-2)) +
  Dist[(n-2)/(n-1),Int[x*Sech[a+b*x]^(n-2),x]] /;
FreeQ[{a,b},x] && RationalQ[n] && n>1 && n≠2
```

- **Reference: G&R 2.643.1h, CRC 429h**

```
Int[x_*Csch[a_.+b_.*x_]^n_,x_Symbol] :=
  -x*Coth[a+b*x]*Csch[a+b*x]^(n-2)/(b*(n-1)) -
  Csch[a+b*x]^(n-2)/(b^2*(n-1)*(n-2)) -
  Dist[(n-2)/(n-1),Int[x*Csch[a+b*x]^(n-2),x]] /;
FreeQ[{a,b},x] && RationalQ[n] && n>1 && n≠2
```

- **Reference: G&R 2.643.2h**
- **Rule: If $m > 1 \land n > 1 \land n \neq 2$, then**

$$\int x^m\,\text{Sech}[a+b\,x]^n\,dx \longrightarrow \frac{x^m\,\text{Tanh}[a+b\,x]\,\text{Sech}[a+b\,x]^{n-2}}{b\,(n-1)} + \frac{m\,x^{m-1}\,\text{Sech}[a+b\,x]^{n-2}}{b^2\,(n-1)\,(n-2)} + \frac{n-2}{n-1}\int x^m\,\text{Sech}[a+b\,x]^{n-2}\,dx - \frac{m\,(m-1)}{b^2\,(n-1)\,(n-2)}\int x^{m-2}\,\text{Sech}[a+b\,x]^{n-2}\,dx$$

- **Program code:**

```
Int[x_^m_*Sech[a_.+b_.*x_]^n_,x_Symbol] :=
  x^m*Tanh[a+b*x]*Sech[a+b*x]^(n-2)/(b*(n-1)) +
  m*x^(m-1)*Sech[a+b*x]^(n-2)/(b^2*(n-1)*(n-2)) +
  Dist[(n-2)/(n-1),Int[x^m*Sech[a+b*x]^(n-2),x]] -
  Dist[m*(m-1)/(b^2*(n-1)*(n-2)),Int[x^(m-2)*Sech[a+b*x]^(n-2),x]] /;
FreeQ[{a,b},x] && RationalQ[{m,n}] && m>1 && n>1 && n≠2
```

- **Reference: G&R 2.643.1h**

```
Int[x_^m_*Csch[a_.+b_.*x_]^n_,x_Symbol] :=
  -x^m*Coth[a+b*x]*Csch[a+b*x]^(n-2)/(b*(n-1)) -
  m*x^(m-1)*Csch[a+b*x]^(n-2)/(b^2*(n-1)*(n-2)) -
  Dist[(n-2)/(n-1),Int[x^m*Csch[a+b*x]^(n-2),x]] +
  Dist[m*(m-1)/(b^2*(n-1)*(n-2)),Int[x^(m-2)*Csch[a+b*x]^(n-2),x]] /;
FreeQ[{a,b},x] && RationalQ[{m,n}] && m>1 && n>1 && n≠2
```

- **Reference: G&R 2.631.3h**
- **Rule: If $n < -1$, then**

$$\int x\,\mathrm{Sech}[a+b\,x]^n\,dx \longrightarrow -\frac{\mathrm{Sech}[a+b\,x]^n}{b^2\,n^2}-\frac{x\,\mathrm{Sinh}[a+b\,x]\,\mathrm{Sech}[a+b\,x]^{n+1}}{b\,n}+\frac{n+1}{n}\int x\,\mathrm{Sech}[a+b\,x]^{n+2}\,dx$$

- **Program code:**

```
Int[x_*Sech[a_.+b_.*x_]^n_,x_Symbol] :=
  -Sech[a+b*x]^n/(b^2*n^2) -
  x*Sinh[a+b*x]*Sech[a+b*x]^(n+1)/(b*n) +
  Dist[(n+1)/n,Int[x*Sech[a+b*x]^(n+2),x]] /;
FreeQ[{a,b},x] && RationalQ[n] && n<-1
```

- **Reference: G&R 2.631.2h**

```
Int[x_*Csch[a_.+b_.*x_]^n_,x_Symbol] :=
  -Csch[a+b*x]^n/(b^2*n^2) -
  x*Cosh[a+b*x]*Csch[a+b*x]^(n+1)/(b*n) -
  Dist[(n+1)/n,Int[x*Csch[a+b*x]^(n+2),x]] /;
FreeQ[{a,b},x] && RationalQ[n] && n<-1
```

- **Reference: G&R 2.631.3h**
- **Rule: If $m > 1 \land n < -1$, then**

$$\int x^m\,\mathrm{Sech}[a+b\,x]^n\,dx \longrightarrow -\frac{m\,x^{m-1}\,\mathrm{Sech}[a+b\,x]^n}{b^2\,n^2}-\frac{x^m\,\mathrm{Sinh}[a+b\,x]\,\mathrm{Sech}[a+b\,x]^{n+1}}{b\,n}+\frac{n+1}{n}\int x^m\,\mathrm{Sech}[a+b\,x]^{n+2}\,dx+\frac{m\,(m-1)}{b^2\,n^2}\int x^{m-2}\,\mathrm{Sech}[a+b\,x]^n\,dx$$

- **Program code:**

```
Int[x_^m_*Sech[a_.+b_.*x_]^n_,x_Symbol] :=
  -m*x^(m-1)*Sech[a+b*x]^n/(b^2*n^2) -
  x^m*Sinh[a+b*x]*Sech[a+b*x]^(n+1)/(b*n) +
  Dist[(n+1)/n,Int[x^m*Sech[a+b*x]^(n+2),x]] +
  Dist[m*(m-1)/(b^2*n^2),Int[x^(m-2)*Sech[a+b*x]^n,x]] /;
FreeQ[{a,b},x] && RationalQ[{m,n}] && m>1 && n<-1
```

- **Reference: G&R 2.631.2h**

```
Int[x_^m_*Csch[a_.+b_.*x_]^n_,x_Symbol] :=
  -m*x^(m-1)*Csch[a+b*x]^n/(b^2*n^2) -
  x^m*Cosh[a+b*x]*Csch[a+b*x]^(n+1)/(b*n) -
  Dist[(n+1)/n,Int[x^m*Csch[a+b*x]^(n+2),x]] +
  Dist[m*(m-1)/(b^2*n^2),Int[x^(m-2)*Csch[a+b*x]^n,x]] /;
FreeQ[{a,b},x] && RationalQ[{m,n}] && m>1 && n<-1
```

$$\int (a + b\,\mathrm{Sech}[c + d\,x]^n)^m\,dx$$

- **Derivation: Algebraic simplification**
- **Basis: If** $n \in \mathbb{Z}$**, then** $a + b\,\mathrm{Sech}[z]^n = \frac{b+a\,\mathrm{Cosh}[z]^n}{\mathrm{Cosh}[z]^n}$
- **Rule: If** $m,\ n \in \mathbb{Z}\ \wedge\ m < 0\ \wedge\ n > 0$**, then**

$$\int (a + b\,\mathrm{Sech}[v]^n)^m\,dx \longrightarrow \int \frac{(b + a\,\mathrm{Cosh}[v]^n)^m}{\mathrm{Cosh}[v]^{m\,n}}\,dx$$

- **Program code:**

```
Int[(a_+b_.*Sech[v_]^n_.)^m_,x_Symbol] :=
  Int[(b+a*Cosh[v]^n)^m/Cosh[v]^(m*n),x] /;
FreeQ[{a,b},x] && IntegersQ[m,n] && m<0 && n>0
```

```
Int[(a_+b_.*Csch[v_]^n_.)^m_,x_Symbol] :=
  Int[(b+a*Sinh[v]^n)^m/Sinh[v]^(m*n),x] /;
FreeQ[{a,b},x] && IntegersQ[m,n] && m<0 && n>0
```

- **Derivation: Algebraic simplification**
- **Basis: If** $n \in \mathbb{Z}$**, then** $a + b\,\mathrm{Sech}[z]^n = \frac{b+a\,\mathrm{Cosh}[z]^n}{\mathrm{Cosh}[z]^n}$
- **Rule: If** $m,\ n,\ p \in \mathbb{Z}\ \wedge\ m < 0\ \wedge\ n > 0$**, then**

$$\int \mathrm{Cosh}[v]^p\,(a + b\,\mathrm{Sech}[v]^n)^m\,dx \longrightarrow \int \mathrm{Cosh}[v]^{p-m\,n}\,(b + a\,\mathrm{Cosh}[v]^n)^m\,dx$$

- **Program code:**

```
Int[Cosh[v_]^p_.*(a_+b_.*Sech[v_]^n_.)^m_,x_Symbol] :=
  Int[Cosh[v]^(p-m*n)*(b+a*Cosh[v]^n)^m,x] /;
FreeQ[{a,b},x] && IntegersQ[m,n,p] && m<0 && n>0
```

```
Int[Sinh[v_]^p_.*(a_+b_.*Csch[v_]^n_.)^m_,x_Symbol] :=
  Int[Sinh[v]^(p-m*n)*(b+a*Sinh[v]^n)^m,x] /;
FreeQ[{a,b},x] && IntegersQ[m,n,p] && m<0 && n>0
```

$$\int \text{Csch}[a+b\,x]^m\,\text{Sech}[a+b\,x]^n\,dx$$

- **Reference: G&R 2.423.49**

- **Rule: If $b>0$, then**

$$\int \text{Csch}[a+b\,x]\,\text{Sech}[a+b\,x]\,dx \longrightarrow \frac{\text{Log}[\text{Tanh}[a+b\,x]]}{b}$$

- **Program code:**

```
Int[Csch[a_.+b_.*x_]*Sech[a_.+b_.*x_],x_Symbol] :=
  Log[Tanh[a+b*x]]/b /;
FreeQ[{a,b},x] && PosQ[b]
```

- **Rule: If $m+n-2=0 \land n-1\neq 0 \land n>0$, then**

$$\int \text{Csch}[a+b\,x]^m\,\text{Sech}[a+b\,x]^n\,dx \longrightarrow \frac{\text{Csch}[a+b\,x]^{m-1}\,\text{Sech}[a+b\,x]^{n-1}}{b\,(n-1)}$$

- **Program code:**

```
Int[Csch[a_.+b_.*x_]^m_*Sech[a_.+b_.*x_]^n_,x_Symbol] :=
  Csch[a+b*x]^(m-1)*Sech[a+b*x]^(n-1)/(b*(n-1)) /;
FreeQ[{a,b,m,n},x] && ZeroQ[m+n-2] && NonzeroQ[n-1] && PosQ[n]
```

- **Derivation: Integration by substitution**

- **Basis: If $m,\ n,\ \frac{m+n}{2}\in\mathbb{Z}$, then $\text{Csch}[z]^m\,\text{Sech}[z]^n = \frac{\left(1-\text{Tanh}[z]^2\right)^{\frac{m+n}{2}-1}}{\text{Tanh}[z]^m}\,\text{Tanh}'[z]$**

- **Rule: If $m,\ n,\ \frac{m+n}{2}\in\mathbb{Z} \land 0<m\leq n$, then**

$$\int \text{Csch}[a+b\,x]^m\,\text{Sech}[a+b\,x]^n\,dx \longrightarrow \frac{1}{b}\,\text{Subst}\left[\int \frac{\left(1-x^2\right)^{\frac{m+n}{2}-1}}{x^m}\,dx,\ x,\ \text{Tanh}[a+b\,x]\right]$$

- **Program code:**

```
Int[Csch[a_.+b_.*x_]^m_.*Sech[a_.+b_.*x_]^n_,x_Symbol] :=
  Dist[1/b,Subst[Int[Regularize[(1-x^2)^((m+n)/2-1)/x^m,x],x],x,Tanh[a+b*x]]] /;
FreeQ[{a,b},x] && IntegersQ[m,n] && EvenQ[m+n] && 0<m<=n
```

- **Reference: G&R 2.411.4**
- **Rule: If** $m < -1 \wedge n > 1$**, then**

$$\int \operatorname{Csch}[a+b\,x]^m \operatorname{Sech}[a+b\,x]^n\,dx \longrightarrow -\frac{\operatorname{Csch}[a+b\,x]^{m+1}\operatorname{Sech}[a+b\,x]^{n-1}}{b\,(n-1)} - \frac{m+1}{n-1}\int \operatorname{Csch}[a+b\,x]^{m+2}\operatorname{Sech}[a+b\,x]^{n-2}\,dx$$

- **Program code:**

```
Int[Csch[a_.+b_.*x_]^m_*Sech[a_.+b_.*x_]^n_,x_Symbol] :=
  -Csch[a+b*x]^(m+1)*Sech[a+b*x]^(n-1)/(b*(n-1)) -
  Dist[(m+1)/(n-1),Int[Csch[a+b*x]^(m+2)*Sech[a+b*x]^(n-2),x]] /;
FreeQ[{a,b},x] && RationalQ[{m,n}] && m<-1 && n>1
```

- **Reference: G&R 2.411.6, CRC 568b, A&S 4.5.86b**
- **Rule: If** $n > 1 \bigwedge \frac{m+n}{2} \notin \mathbb{Z} \bigwedge \neg\left(\frac{n}{2},\ \frac{m-1}{2} \in \mathbb{Z} \bigwedge m > 1\right)$**, then**

$$\int \operatorname{Csch}[a+b\,x]^m \operatorname{Sech}[a+b\,x]^n\,dx \longrightarrow \frac{\operatorname{Csch}[a+b\,x]^{m-1}\operatorname{Sech}[a+b\,x]^{n-1}}{b\,(n-1)} + \frac{m+n-2}{n-1}\int \operatorname{Csch}[a+b\,x]^{m}\operatorname{Sech}[a+b\,x]^{n-2}\,dx$$

- **Program code:**

```
Int[Csch[a_.+b_.*x_]^m_.*Sech[a_.+b_.*x_]^n_,x_Symbol] :=
  Csch[a+b*x]^(m-1)*Sech[a+b*x]^(n-1)/(b*(n-1)) +
  Dist[(m+n-2)/(n-1),Int[Csch[a+b*x]^m*Sech[a+b*x]^(n-2),x]] /;
FreeQ[{a,b,m},x] && RationalQ[n] && n>1 && Not[EvenQ[m+n]] && Not[EvenQ[n] && OddQ[m] && m>1]
```

- **Reference: G&R 2.411.1, CRC 567a, A&S 4.5.85a**
- **Rule: If** $n < -1 \wedge m + n \neq 0$**, then**

$$\int \operatorname{Csch}[a+b\,x]^m \operatorname{Sech}[a+b\,x]^n\,dx \longrightarrow -\frac{\operatorname{Csch}[a+b\,x]^{m-1}\operatorname{Sech}[a+b\,x]^{n+1}}{b\,(m+n)} + \frac{n+1}{m+n}\int \operatorname{Csch}[a+b\,x]^{m}\operatorname{Sech}[a+b\,x]^{n+2}\,dx$$

- **Program code:**

```
Int[Csch[a_.+b_.*x_]^m_.*Sech[a_.+b_.*x_]^n_,x_Symbol] :=
  -Csch[a+b*x]^(m-1)*Sech[a+b*x]^(n+1)/(b*(m+n)) +
  Dist[(n+1)/(m+n),Int[Csch[a+b*x]^m*Sech[a+b*x]^(n+2),x]] /;
FreeQ[{a,b,m},x] && RationalQ[n] && n<-1 && NonzeroQ[m+n]
```

$$\int \text{Csch}[a+b\,x]^m\,\text{Sech}[a+b\,x]^n\,dx$$

- **Reference: G&R 2.423.49'**
- **Rule: If** $\neg\,(b>0)$, **then**

$$\int \text{Csch}[a+b\,x]\,\text{Sech}[a+b\,x]\,dx \longrightarrow -\frac{\text{Log}[\text{Coth}[a+b\,x]]}{b}$$

- **Program code:**

```
Int[Csch[a_.+b_.*x_]*Sech[a_.+b_.*x_],x_Symbol] :=
  -Log[Coth[a+b*x]]/b /;
FreeQ[{a,b},x] && NegQ[b]
```

- **Rule: If** $m+n-2=0 \bigwedge m-1\neq 0 \bigwedge m>0$, **then**

$$\int \text{Csch}[a+b\,x]^m\,\text{Sech}[a+b\,x]^n\,dx \longrightarrow -\frac{\text{Csch}[a+b\,x]^{m-1}\,\text{Sech}[a+b\,x]^{n-1}}{b\,(m-1)}$$

- **Program code:**

```
Int[Csch[a_.+b_.*x_]^m_*Sech[a_.+b_.*x_]^n_,x_Symbol] :=
  -Csch[a+b*x]^(m-1)*Sech[a+b*x]^(n-1)/(b*(m-1)) /;
FreeQ[{a,b,m,n},x] && ZeroQ[m+n-2] && NonzeroQ[m-1] && PosQ[m]
```

- **Derivation: Integration by substitution**
- **Basis: If** $m,\ n,\ \frac{m+n}{2}\in\mathbb{Z}$, **then** $\text{Csch}[z]^m\,\text{Sech}[z]^n = -\frac{\left(-1+\text{Coth}[z]^2\right)^{\frac{m+n}{2}-1}}{\text{Coth}[z]^n}\,\text{Coth}'[z]$
- **Rule: If** $m,\ n,\ \frac{m+n}{2}\in\mathbb{Z} \bigwedge 0<n<m$, **then**

$$\int \text{Csch}[a+b\,x]^m\,\text{Sech}[a+b\,x]^n\,dx \longrightarrow -\frac{1}{b}\,\text{Subst}\Big[\text{Int}\Big[\frac{\left(-1+x^2\right)^{\frac{m+n}{2}-1}}{x^n},\,x\Big],\,x,\,\text{Coth}[a+b\,x]\Big]$$

- **Program code:**

```
Int[Csch[a_.+b_.*x_]^m_*Sech[a_.+b_.*x_]^n_.,x_Symbol] :=
  Dist[-1/b,Subst[Int[Regularize[(-1+x^2)^((m+n)/2-1)/x^n,x],x],x,Coth[a+b*x]]] /;
FreeQ[{a,b},x] && IntegersQ[m,n] && EvenQ[m+n] && 0<n<m
```

- **Reference: G&R 2.411.3**
- **Rule: If** $m > 1 \wedge n < -1$**, then**

$$\int \text{Csch}[a+b\,x]^m\,\text{Sech}[a+b\,x]^n\,dx \longrightarrow -\frac{\text{Csch}[a+b\,x]^{m-1}\,\text{Sech}[a+b\,x]^{n+1}}{b\,(m-1)} - \frac{n+1}{m-1}\int \text{Csch}[a+b\,x]^{m-2}\,\text{Sech}[a+b\,x]^{n+2}\,dx$$

- **Program code:**

```
Int[Csch[a_.+b_.*x_]^m_*Sech[a_.+b_.*x_]^n_,x_Symbol] :=
  -Csch[a+b*x]^(m-1)*Sech[a+b*x]^(n+1)/(b*(m-1)) -
  Dist[(n+1)/(m-1),Int[Csch[a+b*x]^(m-2)*Sech[a+b*x]^(n+2),x]] /;
FreeQ[{a,b},x] && RationalQ[{m,n}] && m>1 && n<-1
```

- **Reference: G&R 2.411.5, CRC 568a, A&S 4.5.86a**
- **Rule: If** $m > 1 \bigwedge \frac{m+n}{2} \notin \mathbb{Z} \bigwedge \neg\left(\frac{m}{2}, \frac{n-1}{2} \in \mathbb{Z} \bigwedge n > 1\right)$**, then**

$$\int \text{Csch}[a+b\,x]^m\,\text{Sech}[a+b\,x]^n\,dx \longrightarrow -\frac{\text{Csch}[a+b\,x]^{m-1}\,\text{Sech}[a+b\,x]^{n-1}}{b\,(m-1)} - \frac{m+n-2}{m-1}\int \text{Csch}[a+b\,x]^{m-2}\,\text{Sech}[a+b\,x]^{n}\,dx$$

- **Program code:**

```
Int[Csch[a_.+b_.*x_]^m_*Sech[a_.+b_.*x_]^n_.,x_Symbol] :=
  -Csch[a+b*x]^(m-1)*Sech[a+b*x]^(n-1)/(b*(m-1)) -
  Dist[(m+n-2)/(m-1),Int[Csch[a+b*x]^(m-2)*Sech[a+b*x]^n,x]] /;
FreeQ[{a,b,n},x] && RationalQ[m] && m>1 && Not[EvenQ[m+n]] && Not[EvenQ[m] && OddQ[n] && n>1]
```

- **Reference: G&R 2.411.2, CRC 567b, A&S 4.5.85b**

- **Rule: If $m < -1 \land m + n \neq 0$, then**

$$\int \mathrm{Csch}[a + b\,x]^m\,\mathrm{Sech}[a + b\,x]^n\,dx \longrightarrow -\frac{\mathrm{Csch}[a + b\,x]^{m+1}\,\mathrm{Sech}[a + b\,x]^{n-1}}{b\,(m + n)} - \frac{m + 1}{m + n}\int \mathrm{Csch}[a + b\,x]^{m+2}\,\mathrm{Sech}[a + b\,x]^n\,dx$$

- **Program code:**

```
Int[Csch[a_.+b_.*x_]^m_*Sech[a_.+b_.*x_]^n_.,x_Symbol] :=
  -Csch[a+b*x]^(m+1)*Sech[a+b*x]^(n-1)/(b*(m+n)) -
  Dist[(m+1)/(m+n),Int[Csch[a+b*x]^(m+2)*Sech[a+b*x]^n,x]] /;
FreeQ[{a,b,n},x] && RationalQ[m] && m<-1 && NonzeroQ[m+n]
```

$$\int \mathrm{Sech}[a+b\,x]^m\,\mathrm{Tanh}[a+b\,x]^n\,dx$$

- **Derivation: Power rule for integration**
- **Rule:**

$$\int \mathrm{Sech}[a+b\,x]^m\,\mathrm{Tanh}[a+b\,x]\,dx \longrightarrow -\frac{\mathrm{Sech}[a+b\,x]^m}{b\,m}$$

- **Program code:**

```
Int[Sech[a_.+b_.*x_]^m_.*Tanh[a_.+b_.*x_]^n_.,x_Symbol] :=
  -Sech[a+b*x]^m/(b*m) /;
FreeQ[{a,b,m},x] && n===1
```

```
Int[Csch[a_.+b_.*x_]^m_.*Coth[a_.+b_.*x_]^n_.,x_Symbol] :=
  -Csch[a+b*x]^m/(b*m) /;
FreeQ[{a,b,m},x] && n===1
```

- **Derivation: Integration by substitution**
- **Basis: If** $\frac{m}{2} \in \mathbb{Z}$, **then** $\mathrm{Sech}[z]^m = \left(1-\mathrm{Tanh}[z]^2\right)^{\frac{m-2}{2}}\,\mathrm{Tanh}'[z]$
- **Rule: If** $\frac{m}{2} \in \mathbb{Z} \bigwedge m>2 \bigwedge \neg\left(\frac{n-1}{2} \in \mathbb{Z} \bigwedge 0<n<m-1\right)$, **then**

$$\int \mathrm{Sech}[a+b\,x]^m\,\mathrm{Tanh}[a+b\,x]^n\,dx \longrightarrow \frac{1}{b}\,\mathrm{Subst}\left[\mathrm{Int}\left[x^n\,\left(1-x^2\right)^{\frac{m-2}{2}},\,x\right],\,x,\,\mathrm{Tanh}[a+b\,x]\right]$$

- **Program code:**

```
Int[Sech[a_.+b_.*x_]^m_*Tanh[a_.+b_.*x_]^n_.,x_Symbol] :=
  Dist[1/b,Subst[Int[Regularize[x^n*(1-x^2)^((m-2)/2),x],x],x,Tanh[a+b*x]]] /;
FreeQ[{a,b,n},x] && EvenQ[m] && m>2 && Not[OddQ[n] && 0<n<m-1]
```

```
Int[Csch[a_.+b_.*x_]^m_*Coth[a_.+b_.*x_]^n_.,x_Symbol] :=
  Dist[-1/b,Subst[Int[Regularize[x^n*(-1+x^2)^((m-2)/2),x],x],x,Coth[a+b*x]]] /;
FreeQ[{a,b,n},x] && EvenQ[m] && m>2 && Not[OddQ[n] && 0<n<m-1]
```

- **Derivation: Integration by substitution**

- **Basis: If** $\frac{n-1}{2} \in \mathbb{Z}$, **then** $\text{Sech}[z]^m \, \text{Tanh}[z]^n = -\text{Sech}[z]^{m-1} \left(1 - \text{Sech}[z]^2\right)^{\frac{n-1}{2}} \text{Sech}'[z]$

- **Rule: If** $\frac{n-1}{2} \in \mathbb{Z} \bigwedge \neg \left(\frac{m}{2} \in \mathbb{Z} \bigwedge 0 < m \leq n+1\right)$, **then**

$$\int \text{Sech}[a + b\,x]^m \, \text{Tanh}[a + b\,x]^n \, dx \longrightarrow -\frac{1}{b} \, \text{Subst}\left[\int x^{m-1} \left(1 - x^2\right)^{\frac{n-1}{2}} dx, \, x, \, \text{Sech}[a + b\,x]\right]$$

- **Program code:**

```
Int[Sech[a_.+b_.*x_]^m_.*Tanh[a_.+b_.*x_]^n_.,x_Symbol] :=
  Dist[-1/b,Subst[Int[Regularize[x^(m-1)*(1-x^2)^((n-1)/2),x],x],x,Sech[a+b*x]]] /;
FreeQ[{a,b,m},x] && OddQ[n] && Not[EvenQ[m] && 0<m<=n+1]
```

```
Int[Csch[a_.+b_.*x_]^m_.*Coth[a_.+b_.*x_]^n_.,x_Symbol] :=
  Dist[-1/b,Subst[Int[Regularize[x^(m-1)*(1+x^2)^((n-1)/2),x],x],x,Csch[a+b*x]]] /;
FreeQ[{a,b,m},x] && OddQ[n] && Not[EvenQ[m] && 0<m<=n+1]
```

- **Reference: G&R 2.411.5, CRC 568a**

- **Rule: If** $m > 1 \bigwedge n < -1 \bigwedge \frac{m}{2} \notin \mathbb{Z}$, **then**

$$\int \text{Sech}[a + b\,x]^m \, \text{Tanh}[a + b\,x]^n \, dx \longrightarrow \frac{\text{Sech}[a + b\,x]^{m-2} \, \text{Tanh}[a + b\,x]^{n+1}}{b\,(n+1)} + \frac{m-2}{n+1} \int \text{Sech}[a + b\,x]^{m-2} \, \text{Tanh}[a + b\,x]^{n+2} \, dx$$

- **Program code:**

```
Int[Sech[a_.+b_.*x_]^m_*Tanh[a_.+b_.*x_]^n_,x_Symbol] :=
  Sech[a+b*x]^(m-2)*Tanh[a+b*x]^(n+1)/(b*(n+1)) +
  Dist[(m-2)/(n+1),Int[Sech[a+b*x]^(m-2)*Tanh[a+b*x]^(n+2),x]] /;
FreeQ[{a,b},x] && RationalQ[{m,n}] && m>1 && n<-1 && Not[EvenQ[m]]
```

- **Reference: G&R 2.411.6, CRC 568b**

```
Int[Csch[a_.+b_.*x_]^m_.*Coth[a_.+b_.*x_]^n_,x_Symbol] :=
  -Csch[a+b*x]^(m-2)*Coth[a+b*x]^(n+1)/(b*(n+1)) -
  Dist[(m-2)/(n+1),Int[Csch[a+b*x]^(m-2)*Coth[a+b*x]^(n+2),x]] /;
FreeQ[{a,b},x] && RationalQ[{m,n}] && m>1 && n<-1 && Not[EvenQ[m]]
```

- **Reference: G&R 2.411.2, CRC 567b**
- **Rule: If** $m < -1 \bigwedge n > 1 \bigwedge \frac{m}{2} \notin \mathbb{Z}$, **then**

$$\int \mathrm{Sech}[a+b\,x]^m\,\mathrm{Tanh}[a+b\,x]^n\,dx \longrightarrow -\frac{\mathrm{Sech}[a+b\,x]^m\,\mathrm{Tanh}[a+b\,x]^{n-1}}{b\,m}+\frac{n-1}{m}\int \mathrm{Sech}[a+b\,x]^{m+2}\,\mathrm{Tanh}[a+b\,x]^{n-2}\,dx$$

- **Program code:**

```
Int[Sech[a_.+b_.*x_]^m_*Tanh[a_.+b_.*x_]^n_,x_Symbol] :=
  -Sech[a+b*x]^m*Tanh[a+b*x]^(n-1)/(b*m) +
  Dist[(n-1)/m,Int[Sech[a+b*x]^(m+2)*Tanh[a+b*x]^(n-2),x]] /;
FreeQ[{a,b},x] && RationalQ[{m,n}] && m<-1 && n>1 && Not[EvenQ[m]]
```

- **Reference: G&R 2.411.1, CRC 567a**

```
Int[Csch[a_.+b_.*x_]^m_*Coth[a_.+b_.*x_]^n_,x_Symbol] :=
  -Csch[a+b*x]^m*Coth[a+b*x]^(n-1)/(b*m) -
  Dist[(n-1)/m,Int[Csch[a+b*x]^(m+2)*Coth[a+b*x]^(n-2),x]] /;
FreeQ[{a,b},x] && RationalQ[{m,n}] && m<-1 && n>1 && Not[EvenQ[m]]
```

- **Reference: G&R 2.411.1, CRC 567a**
- **Rule: If** $m+n+1=0$, **then**

$$\int \mathrm{Sech}[a+b\,x]^m\,\mathrm{Tanh}[a+b\,x]^n\,dx \longrightarrow -\frac{\mathrm{Sech}[a+b\,x]^m\,\mathrm{Tanh}[a+b\,x]^{n+1}}{b\,m}$$

- **Program code:**

```
Int[Sech[a_.+b_.*x_]^m_*Tanh[a_.+b_.*x_]^n_,x_Symbol] :=
  -Sech[a+b*x]^m*Tanh[a+b*x]^(n+1)/(b*m) /;
FreeQ[{a,b,m,n},x] && ZeroQ[m+n+1]
```

- **Reference: G&R 2.411.2, CRC 567b**

```
Int[Csch[a_.+b_.*x_]^m_.*Coth[a_.+b_.*x_]^n_,x_Symbol] :=
  -Csch[a+b*x]^m*Coth[a+b*x]^(n+1)/(b*m) /;
FreeQ[{a,b,m,n},x] && ZeroQ[m+n+1]
```

- **Rule: If** $m < -1 \bigwedge \frac{m}{2} \notin \mathbb{Z}$**, then**

$$\int \mathrm{Sech}[a+b\,x]^m\,\mathrm{Tanh}[a+b\,x]^n\,dx \longrightarrow -\frac{\mathrm{Sech}[a+b\,x]^m\,\mathrm{Tanh}[a+b\,x]^{n+1}}{b\,m}+\frac{m+n+1}{m}\int \mathrm{Sech}[a+b\,x]^{m+2}\,\mathrm{Tanh}[a+b\,x]^n\,dx$$

- **Program code:**

```
Inth[Sech[a_.+b_.*x_]^m_*Tanh[a_.+b_.*x_]^n_,x_Symbol] :=
  -Sech[a+b*x]^m*Tanh[a+b*x]^(n+1)/(b*m) +
  Dist[(m+n+1)/m,Int[Sech[a+b*x]^(m+2)*Tanh[a+b*x]^n,x]] /;
FreeQ[{a,b,n},x] && RationalQ[m] && m<-1 && Not[EvenQ[m]]
```

```
Int[Csch[a_.+b_.*x_]^m_*Coth[a_.+b_.*x_]^n_,x_Symbol] :=
  -Csch[a+b*x]^m*Coth[a+b*x]^(n+1)/(b*m) -
  Dist[(m+n+1)/m,Int[Csch[a+b*x]^(m+2)*Coth[a+b*x]^n,x]] /;
FreeQ[{a,b,n},x] && RationalQ[m] && m<-1 && Not[EvenQ[m]]
```

- **Reference: G&R 2.411.6, CRC 568b**

- **Rule: If** $m > 1 \bigwedge m+n-1 \neq 0 \bigwedge \frac{m}{2} \notin \mathbb{Z} \bigwedge \frac{n-1}{2} \notin \mathbb{Z}$**, then**

$$\int \mathrm{Sech}[a+b\,x]^m\,\mathrm{Tanh}[a+b\,x]^n\,dx \longrightarrow \frac{\mathrm{Sech}[a+b\,x]^{m-2}\,\mathrm{Tanh}[a+b\,x]^{n+1}}{b\,(m+n-1)}+\frac{m-2}{m+n-1}\int \mathrm{Sech}[a+b\,x]^{m-2}\,\mathrm{Tanh}[a+b\,x]^n\,dx$$

- **Program code:**

```
Int[Sech[a_.+b_.*x_]^m_*Tanh[a_.+b_.*x_]^n_,x_Symbol] :=
  Sech[a+b*x]^(m-2)*Tanh[a+b*x]^(n+1)/(b*(m+n-1)) +
  Dist[(m-2)/(m+n-1),Int[Sech[a+b*x]^(m-2)*Tanh[a+b*x]^n,x]] /;
FreeQ[{a,b,n},x] && RationalQ[m] && m>1 && NonzeroQ[m+n-1] && Not[EvenQ[m]] && Not[OddQ[n]]
```

- **Reference: G&R 2.411.5, CRC 568a**

```
Int[Csch[a_.+b_.*x_]^m_*Coth[a_.+b_.*x_]^n_,x_Symbol] :=
  -Csch[a+b*x]^(m-2)*Coth[a+b*x]^(n+1)/(b*(m+n-1)) -
  Dist[(m-2)/(m+n-1),Int[Csch[a+b*x]^(m-2)*Coth[a+b*x]^n,x]] /;
FreeQ[{a,b,n},x] && RationalQ[m] && m>1 && NonzeroQ[m+n-1] && Not[EvenQ[m]] && Not[OddQ[n]]
```

- **Reference: G&R 2.411.3**
- **Rule: If** $n > 1 \bigwedge m + n - 1 \neq 0 \bigwedge \frac{m}{2} \notin \mathbb{Z} \bigwedge \frac{n-1}{2} \notin \mathbb{Z}$**, then**

$$\int \text{Sech}[a + b\,x]^m\,\text{Tanh}[a + b\,x]^n\,dx \longrightarrow -\frac{\text{Sech}[a + b\,x]^m\,\text{Tanh}[a + b\,x]^{n-1}}{b\,(m + n - 1)} + \frac{n - 1}{m + n - 1}\int \text{Sech}[a + b\,x]^m\,\text{Tanh}[a + b\,x]^{n-2}\,dx$$

- **Program code:**

```
Int[Sech[a_.+b_.*x_]^m_.*Tanh[a_.+b_.*x_]^n_,x_Symbol] :=
  -Sech[a+b*x]^m*Tanh[a+b*x]^(n-1)/(b*(m+n-1)) +
  Dist[(n-1)/(m+n-1),Int[Sech[a+b*x]^m*Tanh[a+b*x]^(n-2),x]] /;
FreeQ[{a,b,m},x] && RationalQ[n] && n>1 && NonzeroQ[m+n-1] && Not[EvenQ[m]] && Not[OddQ[n]]
```

- **Reference: G&R 2.411.4**

```
Int[Csch[a_.+b_.*x_]^m_.*Coth[a_.+b_.*x_]^n_,x_Symbol] :=
  -Csch[a+b*x]^m*Coth[a+b*x]^(n-1)/(b*(m+n-1)) +
  Dist[(n-1)/(m+n-1),Int[Csch[a+b*x]^m*Coth[a+b*x]^(n-2),x]] /;
FreeQ[{a,b,m},x] && RationalQ[n] && n>1 && NonzeroQ[m+n-1] && Not[EvenQ[m]] && Not[OddQ[n]]
```

- **Reference: G&R 2.411.4**
- **Rule: If** $n < -1 \bigwedge \frac{m}{2} \notin \mathbb{Z}$**, then**

$$\int \text{Sech}[a + b\,x]^m\,\text{Tanh}[a + b\,x]^n\,dx \longrightarrow \frac{\text{Sech}[a + b\,x]^m\,\text{Tanh}[a + b\,x]^{n+1}}{b\,(n + 1)} + \frac{m + n + 1}{n + 1}\int \text{Sech}[a + b\,x]^m\,\text{Tanh}[a + b\,x]^{n+2}\,dx$$

- **Program code:**

```
Int[Sech[a_.+b_.*x_]^m_*Tanh[a_.+b_.*x_]^n_,x_Symbol] :=
  Sech[a+b*x]^m*Tanh[a+b*x]^(n+1)/(b*(n+1)) +
  Dist[(m+n+1)/(n+1),Int[Sech[a+b*x]^m*Tanh[a+b*x]^(n+2),x]] /;
FreeQ[{a,b,m},x] && RationalQ[n] && n<-1 && Not[EvenQ[m]]
```

- **Reference: G&R 2.411.3**

```
Int[Csch[a_.+b_.*x_]^m_.*Coth[a_.+b_.*x_]^n_,x_Symbol] :=
  Csch[a+b*x]^m*Coth[a+b*x]^(n+1)/(b*(n+1)) +
  Dist[(m+n+1)/(n+1),Int[Csch[a+b*x]^m*Coth[a+b*x]^(n+2),x]] /;
FreeQ[{a,b,m},x] && RationalQ[n] && n<-1 && Not[EvenQ[m]]
```

$$\int x^m \operatorname{Sech}[a + b x^n]^p \operatorname{Sinh}[a + b x^n] \, dx$$

- **Derivation: Integration by parts**

- **Rule: If** $n \in \mathbb{Z} \wedge m - n \geq 0 \wedge p - 1 \neq 0$**, then**

$$\int x^m \operatorname{Sech}[a + b x^n]^p \operatorname{Sinh}[a + b x^n] \, dx \longrightarrow -\frac{x^{m-n+1} \operatorname{Sech}[a + b x^n]^{p-1}}{b n (p-1)} + \frac{m-n+1}{b n (p-1)} \int x^{m-n} \operatorname{Sech}[a + b x^n]^{p-1} \, dx$$

- **Program code:**

```
Int[x_^m_.*Sech[a_.+b_.*x_^n_.]^p_*Sinh[a_.+b_.*x_^n_.],x_Symbol] :=
  -x^(m-n+1)*Sech[a+b*x^n]^(p-1)/(b*n*(p-1)) +
  Dist[(m-n+1)/(b*n*(p-1)),Int[x^(m-n)*Sech[a+b*x^n]^(p-1),x]] /;
FreeQ[{a,b,p},x] && RationalQ[m] && IntegerQ[n] && m-n>=0 && NonzeroQ[p-1]
```

```
Int[x_^m_.*Csch[a_.+b_.*x_^n_.]^p_*Cosh[a_.+b_.*x_^n_.],x_Symbol] :=
  -x^(m-n+1)*Csch[a+b*x^n]^(p-1)/(b*n*(p-1)) +
  Dist[(m-n+1)/(b*n*(p-1)),Int[x^(m-n)*Csch[a+b*x^n]^(p-1),x]] /;
FreeQ[{a,b,p},x] && RationalQ[m] && IntegerQ[n] && m-n>=0 && NonzeroQ[p-1]
```

$$\int x^m \operatorname{Sech}[a+b\,x^n]^p \operatorname{Tanh}[a+b\,x^n]\,dx$$

- **Derivation: Integration by parts**

- **Note: Dummy exponent $q = 1$ required in program code so InputForm of integrand is recognized.**

- **Rule: If $n \in \mathbb{Z} \land m-n \geq 0$, then**

$$\int x^m \operatorname{Sech}[a+b\,x^n]^p \operatorname{Tanh}[a+b\,x^n]\,dx \longrightarrow -\frac{x^{m-n+1} \operatorname{Sech}[a+b\,x^n]^p}{b\,n\,p}+\frac{m-n+1}{b\,n\,p}\int x^{m-n} \operatorname{Sech}[a+b\,x^n]^p\,dx$$

- **Program code:**

```
Int[x_^m_.*Sech[a_.+b_.*x_^n_.]^p_.*Tanh[a_.+b_.*x_^n_.]^q_.,x_Symbol] :=
  -x^(m-n+1)*Sech[a+b*x^n]^p/(b*n*p) +
  Dist[(m-n+1)/(b*n*p),Int[x^(m-n)*Sech[a+b*x^n]^p,x]] /;
FreeQ[{a,b,p},x] && RationalQ[m] && IntegerQ[n] && m-n>=0 && q===1
```

```
Int[x_^m_.*Csch[a_.+b_.*x_^n_.]^p_.*Coth[a_.+b_.*x_^n_.]^q_.,x_Symbol] :=
  -x^(m-n+1)*Csch[a+b*x^n]^p/(b*n*p) +
  Dist[(m-n+1)/(b*n*p),Int[x^(m-n)*Csch[a+b*x^n]^p,x]] /;
FreeQ[{a,b,p},x] && RationalQ[m] && IntegerQ[n] && m-n>=0 && q===1
```

$$\int \text{Sech}[a + b\,\text{Log}[c\,x^n]]^p\,dx$$

- **Derivation: Algebraic simplification**
- **Basis:** $\text{Sech}[b\,\text{Log}[c\,x^n]] = \frac{2}{(c\,x^n)^b + \frac{1}{(c\,x^n)^b}}$
- **Rule:**

$$\int \text{Sech}[b\,\text{Log}[c\,x^n]]^p\,dx \longrightarrow \int \left(\frac{2}{(c\,x^n)^b + \frac{1}{(c\,x^n)^b}}\right)^p dx$$

- **Program code:**

```
Int[Sech[b_.*Log[c_.*x_^n_.]]^p_.,x_Symbol] :=
  Int[(2/((c*x^n)^b+1/(c*x^n)^b))^p,x] /;
FreeQ[c,x] && RationalQ[{b,n,p}]
```

```
Int[Csch[b_.*Log[c_.*x_^n_.]]^p_.,x_Symbol] :=
  Int[(2/((c*x^n)^b - 1/(c*x^n)^b))^p,x] /;
FreeQ[c,x] && RationalQ[{b,n,p}]
```

- **Rule: If** $p - 1 \neq 0 \land b^2\,n^2\,(p-2)^2 - 1 = 0$**, then**

$$\int \text{Sech}[a + b\,\text{Log}[c\,x^n]]^p\,dx \longrightarrow$$

$$\frac{x\,\text{Tanh}[a + b\,\text{Log}[c\,x^n]]\,\text{Sech}[a + b\,\text{Log}[c\,x^n]]^{p-2}}{b\,n\,(p-1)} + \frac{x\,\text{Sech}[a + b\,\text{Log}[c\,x^n]]^{p-2}}{b^2\,n^2\,(p-1)\,(p-2)}$$

- **Program code:**

```
Int[Sech[a_.+b_.*Log[c_.*x_^n_.]]^p_,x_Symbol] :=
  x*Tanh[a+b*Log[c*x^n]]*Sech[a+b*Log[c*x^n]]^(p-2)/(b*n*(p-1)) +
  x*Sech[a+b*Log[c*x^n]]^(p-2)/(b^2*n^2*(p-1)*(p-2)) /;
FreeQ[{a,b,c,n,p},x] && NonzeroQ[p-1] && ZeroQ[b^2*n^2*(p-2)^2-1]
```

```
Int[Csch[a_.+b_.*Log[c_.*x_^n_.]]^p_,x_Symbol] :=
  -x*Coth[a+b*Log[c*x^n]]*Csch[a+b*Log[c*x^n]]^(p-2)/(b*n*(p-1)) -
  x*Csch[a+b*Log[c*x^n]]^(p-2)/(b^2*n^2*(p-1)*(p-2)) /;
FreeQ[{a,b,c,n,p},x] && NonzeroQ[p-1] && ZeroQ[b^2*n^2*(p-2)^2-1]
```

- **Rule: If** $p>1 \wedge p\neq 2 \wedge b^2 n^2 (p-2)^2-1\neq 0$**, then**

$$\int \text{Sech}[a+b\,\text{Log}[c\,x^n]]^p\,dx \longrightarrow \frac{x\,\text{Tanh}[a+b\,\text{Log}[c\,x^n]]\,\text{Sech}[a+b\,\text{Log}[c\,x^n]]^{p-2}}{b\,n\,(p-1)}+\frac{x\,\text{Sech}[a+b\,\text{Log}[c\,x^n]]^{p-2}}{b^2 n^2 (p-1)(p-2)}+\frac{b^2 n^2 (p-2)^2-1}{b^2 n^2 (p-1)(p-2)}\int \text{Sech}[a+b\,\text{Log}[c\,x^n]]^{p-2}\,dx$$

- **Program code:**

```
Int[Sech[a_.+b_.*Log[c_.*x_^n_.]]^p_,x_Symbol] :=
  x*Tanh[a+b*Log[c*x^n]]*Sech[a+b*Log[c*x^n]]^(p-2)/(b*n*(p-1)) +
  x*Sech[a+b*Log[c*x^n]]^(p-2)/(b^2*n^2*(p-1)*(p-2)) +
  Dist[(b^2*n^2*(p-2)^2-1)/(b^2*n^2*(p-1)*(p-2)),Int[Sech[a+b*Log[c*x^n]]^(p-2),x]] /;
FreeQ[{a,b,c,n},x] && RationalQ[p] && p>1 && p≠2 && NonzeroQ[b^2*n^2*(p-2)^2-1]
```

```
Int[Csch[a_.+b_.*Log[c_.*x_^n_.]]^p_,x_Symbol] :=
  -x*Coth[a+b*Log[c*x^n]]*Csch[a+b*Log[c*x^n]]^(p-2)/(b*n*(p-1)) -
  x*Csch[a+b*Log[c*x^n]]^(p-2)/(b^2*n^2*(p-1)*(p-2)) -
  Dist[(b^2*n^2*(p-2)^2-1)/(b^2*n^2*(p-1)*(p-2)),Int[Csch[a+b*Log[c*x^n]]^(p-2),x]] /;
FreeQ[{a,b,c,n},x] && RationalQ[p] && p>1 && p≠2 && NonzeroQ[b^2*n^2*(p-2)^2-1]
```

- **Rule: If** $p<-1 \wedge b^2 n^2 p^2-1\neq 0$**, then**

$$\int \text{Sech}[a+b\,\text{Log}[c\,x^n]]^p\,dx \longrightarrow -\frac{b\,n\,p\,x\,\text{Sinh}[a+b\,\text{Log}[c\,x^n]]\,\text{Sech}[a+b\,\text{Log}[c\,x^n]]^{p+1}}{b^2 n^2 p^2-1}-\frac{x\,\text{Sech}[a+b\,\text{Log}[c\,x^n]]^{p}}{b^2 n^2 p^2-1}+\frac{b^2 n^2 p\,(p+1)}{b^2 n^2 p^2-1}\int \text{Sech}[a+b\,\text{Log}[c\,x^n]]^{p+2}\,dx$$

- **Program code:**

```
Int[Sech[a_.+b_.*Log[c_.*x_^n_.]]^p_,x_Symbol] :=
  -b*n*p*x*Sech[a+b*Log[c*x^n]]^(p+1)*Sinh[a+b*Log[c*x^n]]/(b^2*n^2*p^2-1) -
  x*Sech[a+b*Log[c*x^n]]^p/(b^2*n^2*p^2-1) +
  Dist[b^2*n^2*p*(p+1)/(b^2*n^2*p^2-1),Int[Sech[a+b*Log[c*x^n]]^(p+2),x]] /;
FreeQ[{a,b,c,n},x] && RationalQ[p] && p<-1 && NonzeroQ[b^2*n^2*p^2-1]
```

```
Int[Csch[a_.+b_.*Log[c_.*x_^n_.]]^p_,x_Symbol] :=
  -b*n*p*x*Cosh[a+b*Log[c*x^n]]*Csch[a+b*Log[c*x^n]]^(p+1)/(b^2*n^2*p^2-1) -
  x*Csch[a+b*Log[c*x^n]]^p/(b^2*n^2*p^2-1) -
  Dist[b^2*n^2*p*(p+1)/(b^2*n^2*p^2-1),Int[Csch[a+b*Log[c*x^n]]^(p+2),x]] /;
FreeQ[{a,b,c,n},x] && RationalQ[p] && p<-1 && NonzeroQ[b^2*n^2*p^2-1]
```

$$\int x^m \operatorname{Sech}[a + b \operatorname{Log}[c x^n]]^p \, dx$$

- **Derivation: Algebraic simplification**
- **Basis:** $\operatorname{Sech}[b \operatorname{Log}[c x^n]] = \frac{2}{(c x^n)^b + \frac{1}{(c x^n)^b}}$
- **Rule:**

$$\int x^m \operatorname{Sech}[b \operatorname{Log}[c x^n]]^p \, dx \longrightarrow \int x^m \left(\frac{2}{(c x^n)^b + \frac{1}{(c x^n)^b}}\right)^p dx$$

- **Program code:**

```
Int[x_^m_.Sech[b_.*Log[c_.*x_^n_.]]^p_.,x_Symbol] :=
  Int[x^m*(2/((c*x^n)^b+1/(c*x^n)^b))^p,x] /;
FreeQ[c,x] && RationalQ[{b,m,n,p}]
```

```
Int[x_^m_.*Csch[b_.*Log[c_.*x_^n_.]]^p_.,x_Symbol] :=
  Int[x^m*(2/((c*x^n)^b - 1/(c*x^n)^b))^p,x] /;
FreeQ[c,x] && RationalQ[{b,m,n,p}]
```

- **Rule: If** $m + 1 \neq 0 \land p - 1 \neq 0 \land b^2 n^2 (p - 2)^2 + (m + 1)^2 = 0$, **then**

$$\int x^m \operatorname{Sech}[a + b \operatorname{Log}[c x^n]]^p \, dx \longrightarrow \frac{x^{m+1} (b n (p - 2) + (m + 1) \operatorname{Tanh}[a + b \operatorname{Log}[c x^n]]) \operatorname{Sech}[a + b \operatorname{Log}[c x^n]]^{p-2}}{b n (m + 1) (p - 1)}$$

- **Program code:**

```
Int[x_^m_.*Sec[a_.+b_.*Log[c_.*x_^n_.]]^p_,x_Symbol] :=
  x^(m+1)*(b*n*(p-2)+(m+1)*Tan[a+b*Log[c*x^n]])*Sec[a+b*Log[c*x^n]]^(p-2)/(b*n*(m+1)*(p-1)) /;
FreeQ[{a,b,c,m,n,p},x] && NonzeroQ[m+1] && NonzeroQ[p-1] && ZeroQ[b^2*n^2*(p-2)^2+(m+1)^2]
```

```
Int[x_^m_.*Csc[a_.+b_.*Log[c_.*x_^n_.]]^p_,x_Symbol] :=
  x^(m+1)*(b*n*(p-2)-(m+1)*Cot[a+b*Log[c*x^n]])*Csc[a+b*Log[c*x^n]]^(p-2)/(b*n*(m+1)*(p-1)) /;
FreeQ[{a,b,c,m,n,p},x] && NonzeroQ[m+1] && NonzeroQ[p-1] && ZeroQ[b^2*n^2*(p-2)^2+(m+1)^2]
```

- **Rule: If** $p > 1 \land p \neq 2 \land b^2 n^2 (p-2)^2 - (m+1)^2 \neq 0$, **then**

$$\int x^m \operatorname{Sech}[a + b \operatorname{Log}[c x^n]]^p \, dx \longrightarrow \frac{x^{m+1} \operatorname{Tanh}[a + b \operatorname{Log}[c x^n]] \operatorname{Sech}[a + b \operatorname{Log}[c x^n]]^{p-2}}{b n (p-1)} + \frac{(m+1) x^{m+1} \operatorname{Sech}[a + b \operatorname{Log}[c x^n]]^{p-2}}{b^2 n^2 (p-1)(p-2)} + \frac{b^2 n^2 (p-2)^2 - (m+1)^2}{b^2 n^2 (p-1)(p-2)} \int x^m \operatorname{Sech}[a + b \operatorname{Log}[c x^n]]^{p-2} \, dx$$

- **Program code:**

```
Int[x_^m_.*Sech[a_.+b_.*Log[c_.*x_^n_.]]^p_,x_Symbol] :=
  x^(m+1)*Tanh[a+b*Log[c*x^n]]*Sech[a+b*Log[c*x^n]]^(p-2)/(b*n*(p-1)) +
  (m+1)*x^(m+1)*Sech[a+b*Log[c*x^n]]^(p-2)/(b^2*n^2*(p-1)*(p-2)) +
  Dist[(b^2*n^2*(p-2)^2-(m+1)^2)/(b^2*n^2*(p-1)*(p-2)),Int[x^m*Sech[a+b*Log[c*x^n]]^(p-2),x]] /;
FreeQ[{a,b,c,m,n},x] && RationalQ[p] && p>1 && p≠2 && NonzeroQ[b^2*n^2*(p-2)^2-(m+1)^2]
```

```
Int[x_^m_.*Csch[a_.+b_.*Log[c_.*x_^n_.]]^p_,x_Symbol] :=
  -x^(m+1)*Coth[a+b*Log[c*x^n]]*Csch[a+b*Log[c*x^n]]^(p-2)/(b*n*(p-1)) -
  (m+1)*x^(m+1)*Csch[a+b*Log[c*x^n]]^(p-2)/(b^2*n^2*(p-1)*(p-2)) -
  Dist[(b^2*n^2*(p-2)^2-(m+1)^2)/(b^2*n^2*(p-1)*(p-2)),Int[x^m*Csch[a+b*Log[c*x^n]]^(p-2),x]] /;
FreeQ[{a,b,c,m,n},x] && RationalQ[p] && p>1 && p≠2 && NonzeroQ[b^2*n^2*(p-2)^2-(m+1)^2]
```

- **Rule: If** $p < -1 \land b^2 n^2 p^2 - (m+1)^2 \neq 0$, **then**

$$\int x^m \operatorname{Sech}[a + b \operatorname{Log}[c x^n]]^p \, dx \longrightarrow -\frac{b n p x^{m+1} \operatorname{Sinh}[a + b \operatorname{Log}[c x^n]] \operatorname{Sech}[a + b \operatorname{Log}[c x^n]]^{p+1}}{b^2 n^2 p^2 - (m+1)^2} - \frac{(m+1) x^{m+1} \operatorname{Sech}[a + b \operatorname{Log}[c x^n]]^p}{b^2 n^2 p^2 - (m+1)^2} + \frac{b^2 n^2 p (p+1)}{b^2 n^2 p^2 - (m+1)^2} \int x^m \operatorname{Sech}[a + b \operatorname{Log}[c x^n]]^{p+2} \, dx$$

- **Program code:**

```
Int[x_^m_.*Sech[a_.+b_.*Log[c_.*x_^n_.]]^p_,x_Symbol] :=
  -(m+1)*x^(m+1)*Sech[a+b*Log[c*x^n]]^p/(b^2*n^2*p^2-(m+1)^2) -
  b*n*p*x^(m+1)*Sech[a+b*Log[c*x^n]]^(p+1)*Sinh[a+b*Log[c*x^n]]/(b^2*n^2*p^2-(m+1)^2) +
  Dist[b^2*n^2*p*(p+1)/(b^2*n^2*p^2-(m+1)^2),Int[x^m*Sech[a+b*Log[c*x^n]]^(p+2),x]] /;
FreeQ[{a,b,c,m,n},x] && RationalQ[p] && p<-1 && NonzeroQ[b^2*n^2*p^2-(m+1)^2]
```

```
Int[x_^m_.*Csch[a_.+b_.*Log[c_.*x_^n_.]]^p_,x_Symbol] :=
  -(m+1)*x^(m+1)*Csch[a+b*Log[c*x^n]]^p/(b^2*n^2*p^2-(m+1)^2) -
  b*n*p*x^(m+1)*Cosh[a+b*Log[c*x^n]]*Csch[a+b*Log[c*x^n]]^(p+1)/(b^2*n^2*p^2-(m+1)^2) -
  Dist[b^2*n^2*p*(p+1)/(b^2*n^2*p^2-(m+1)^2),Int[x^m*Csch[a+b*Log[c*x^n]]^(p+2),x]] /;
FreeQ[{a,b,c,m,n},x] && RationalQ[p] && p<-1 && NonzeroQ[b^2*n^2*p^2-(m+1)^2]
```