Integration Rules for

$$\int \left(\sin^j(z)\right)^m \left(A + B \sin^k(z)\right) \left(a + b \sin^k(z)\right)^n \, dz \text{ when } j^2 = 1 \bigwedge k^2 = 1$$

Domain Map



Legend:

- The rule number in a colored region indicates the rule to use for integrals in that region.
- The rule number next to a colored line indicates the rule to use for integrals on that line.
- A white region or line indicates there is no rule for integrals in that region or on that line.
- A solid black line indicates integrals on that line are handled by rules in another section.
- A dashed black line on the border of a region indicates integrals on that border are handled by the rule for that region.
- The arrow(s) following a rule number indicates the direction the rule drives integrands in the n×m exponent plane.
- A ⋄ following a rule number indicates the rule transforms the integrand into a form handled by another section.
- A red (stop) disk indicates the terminal rule to use for the point at the center of the disk.
- A cyan disk indicates the non-terminal rule to use for the point at the center of the disk.

Integration Rules for

$\int \left(\sin^j(z)\right)^m \left(A + B \sin^k(z)\right) dz$ when $j^2 = 1 \bigwedge k^2 = 1$

Rule a: $\int \mathrm{Sin}[c + d\,x]^j \left(A + B\,\mathrm{Sin}[c + d\,x]^{-j}\right) dx$

- **Derivation: Algebraic expansion**
- **Rule a: If** $j^2 = 1$**, then**

$$\int \mathrm{Sin}[c + d\,x]^j \left(A + B\,\mathrm{Sin}[c + d\,x]^{-j}\right) dx \longrightarrow B\,x + A \int \mathrm{Sin}[c + d\,x]^j\, dx$$

- **Program code:**

```
Int[sin[c_.+d_.*x_]^j_.*(A_+B_.*sin[c_.+d_.*x_]^k_.),x_Symbol] :=
  B*x + Dist[A,Int[sin[c+d*x]^j,x]] /;
FreeQ[{c,d,A,B},x] && OneQ[j^2] && ZeroQ[j+k]
```

Rule b: $\int \left(\text{Sin}[c+d\,x]^j\right)^{m/2}\left(A+B\,\text{Sin}[c+d\,x]^{-j}\right)\,dx$

- **Derivation: Algebraic expansion**
- **Rule b1:**

$$\int\sqrt{\text{Sin}[c+d\,x]}\,(A+B\,\text{Csc}[c+d\,x])\,dx\longrightarrow A\int\sqrt{\text{Sin}[c+d\,x]}\,dx+B\int\frac{1}{\sqrt{\text{Sin}[c+d\,x]}}\,dx$$

- **Program code:**

```
Int[Sqrt[sin[c_.+d_.*x_]]*(A_+B_.*sin[c_.+d_.*x_]^(-1)),x_Symbol] :=
  Dist[A,Int[Sqrt[sin[c+d*x]],x]] +
  Dist[B,Int[1/Sqrt[sin[c+d*x]],x]] /;
FreeQ[{c,d,A,B},x]
```

- **Derivation: Piecewise constant extraction**
- **Basis:** $\partial_z\left(f[z]^m\,(1/f[z])^m\right)=0$
- **Note:** For some strange reason, *Mathematica* overly agressively evaluates $\frac{1}{\sqrt{f[z]}\,\sqrt{1/f[z]}}$ to $\sqrt{f[z]}\,\sqrt{1/f[z]}$.
- **Rule b2:** If $m^2=1$, then

$$\int\text{Csc}[c+d\,x]^{m/2}\,(A+B\,\text{Sin}[c+d\,x])\,dx\longrightarrow\text{Sin}[c+d\,x]^{m/2}\,\text{Csc}[c+d\,x]^{m/2}\int\frac{A+B\,\text{Sin}[c+d\,x]}{\text{Sin}[c+d\,x]^{m/2}}\,dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^(-1))^m_*(A_.+B_.*sin[c_.+d_.*x_]),x_Symbol] :=
  Dist[Sin[c+d*x]^m*Csc[c+d*x]^m,Int[(A+B*sin[c+d*x])/sin[c+d*x]^m,x]] /;
FreeQ[{c,d,A,B},x] && ZeroQ[m^2-1/4]
```

Rules 7 – 8: $\int \left(\text{Sin}[c+d\,x]^j\right)^m \left(A+B\,\text{Sin}[c+d\,x]^k\right) dx$

- **Derivation: Rule 5 with $a = 1$, $b = 0$ and $n = 0$**
- **Rule 7: If $j^2 = k^2 = 1 \land j\,k\,m < -1$, then**

$$\int \left(\text{Sin}[c+d\,x]^j\right)^m \left(A+B\,\text{Sin}[c+d\,x]^k\right) dx \longrightarrow \frac{A\,\text{Cos}[c+d\,x]\,\left(\text{Sin}[c+d\,x]^j\right)^{m+j\,k}}{d\left(j\,k\,m+\frac{k+1}{2}\right)} + \frac{1}{j\,k\,m+\frac{k+1}{2}} \int \left(\text{Sin}[c+d\,x]^j\right)^{m+j\,k} \left(B\left(j\,k\,m+\frac{k+1}{2}\right)+A\left(j\,k\,m+\frac{k+3}{2}\right)\text{Sin}[c+d\,x]^k\right) dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_*(A_+B_.*sin[c_.+d_.*x_]^k_.),x_Symbol] :=
  A*Cos[c+d*x]*(Sin[c+d*x]^j)^(m+j*k)/(d*(j*k*m+(k+1)/2)) +
  Dist[1/(j*k*m+(k+1)/2),
    Int[(sin[c+d*x]^j)^(m+j*k)*Sim[B*(j*k*m+(k+1)/2)+A*(j*k*m+(k+3)/2)*sin[c+d*x]^k,x],x]] /;
FreeQ[{c,d,A,B},x] && OneQ[j^2,k^2] && RationalQ[m] && j*k*m<-1
```

- **Derivation: Rule 3b with $n = 0$**
- **Rule 8: If $j^2 = k^2 = 1 \land j\,k\,m > 0 \land m^2 \neq 1$, then**

$$\int \left(\text{Sin}[c+d\,x]^j\right)^m \left(A+B\,\text{Sin}[c+d\,x]^k\right) dx \longrightarrow -\frac{B\,\text{Cos}[c+d\,x]\,\left(\text{Sin}[c+d\,x]^j\right)^{m}}{d\left(j\,k\,m+\frac{k+1}{2}\right)} + \frac{1}{j\,k\,m+\frac{k+1}{2}} \int \left(\text{Sin}[c+d\,x]^j\right)^{m-j\,k} \left(B\left(j\,k\,m+\frac{k-1}{2}\right)+A\left(j\,k\,m+\frac{k+1}{2}\right)\text{Sin}[c+d\,x]^k\right) dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_*(A_+B_.*sin[c_.+d_.*x_]^k_.),x_Symbol] :=
  -B*Cos[c+d*x]*(Sin[c+d*x]^j)^m/(d*(j*k*m+(k+1)/2)) +
  Dist[1/(j*k*m+(k+1)/2),
    Int[(sin[c+d*x]^j)^(m-j*k)*(B*(j*k*m+(k-1)/2)+A*(j*k*m+(k+1)/2)*sin[c+d*x]^k),x]] /;
FreeQ[{c,d,A,B},x] && OneQ[j^2,k^2] && RationalQ[m] && j*k*m>0 && m^2≠1
```

Integration Rules for

$\int\left(A+B\sin^k(z)\right)\left(a+b\sin^k(z)\right)^n\,dz$ when $k^2=1$

Rule c: $\int\left(A+B\,\mathrm{Sin}[c+d\,x]^k\right)\left(a+b\,\mathrm{Sin}[c+d\,x]^k\right)dx$

- **Derivation: Rule 3a with $m=0$ and $n=1$**
- **Rule c: If $k^2=1$, then**

$$\int\left(A+B\,\mathrm{Sin}[c+d\,x]^k\right)\left(a+b\,\mathrm{Sin}[c+d\,x]^k\right)dx\longrightarrow$$
$$\frac{(4\,a\,A+b\,B\,(k+1))\,x}{4}-\frac{2\,b\,B\,\mathrm{Cos}[c+d\,x]\,\mathrm{Sin}[c+d\,x]^k}{d\,(k+3)}+(b\,A+a\,B)\int\mathrm{Sin}[c+d\,x]^k\,dx$$

- **Program code:**

```
Int[(A_.+B_.*sin[c_.+d_.*x_]^k_.)*(a_+b_.*sin[c_.+d_.*x_]^k_.),x_Symbol] :=
  (4*a*A+b*B*(k+1))*x/4 - (2*b*B*Cos[c+d*x]*Sin[c+d*x]^k)/(d*(k+3)) + (b*A+a*B)*Int[sin[c+d*x]^k,x]
FreeQ[{a,b,c,d,A,B},x] && OneQ[k^2]
```

Rule d: $\int \frac{A + B \operatorname{Sin}[c + d x]^k}{a + b \operatorname{Sin}[c + d x]^k} \, dx$

- **Derivation: Algebraic simplification**
- **Basis: If $b A - a B = 0$, then $\frac{A+B z}{a+b z} = \frac{B}{b}$**
- **Rule d1: If $k^2 = 1 \land b A - a B = 0$, then**

$$\int \frac{A + B \operatorname{Sin}[c + d x]^k}{a + b \operatorname{Sin}[c + d x]^k} \, dx \longrightarrow \frac{B x}{b}$$

- **Program code:**

```
Int[(A_+B_.*sin[c_.+d_.*x_]^k_.)/(a_+b_.*sin[c_.+d_.*x_]^k_.),x_Symbol] :=
  B*x/b /;
FreeQ[{a,b,c,d,A,B},x] && OneQ[k^2] && ZeroQ[b*A-a*B]
```

- **Reference: G&R 2.551.2**
- **Derivation: Algebraic expansion**
- **Basis: $\frac{A+B z}{a+b z} = \frac{B}{b} + \frac{b A-a B}{b (a+b z)}$**
- **Rule d2: If $a^2 - b^2 \neq 0 \land b A - a B \neq 0$, then**

$$\int \frac{A + B \operatorname{Sin}[c + d x]}{a + b \operatorname{Sin}[c + d x]} \, dx \longrightarrow \frac{B x}{b} + \frac{b A - a B}{b} \int \frac{1}{a + b \operatorname{Sin}[c + d x]} \, dx$$

- **Program code:**

```
Int[(A_.+B_.*sin[c_.+d_.*x_])/(a_+b_.*sin[c_.+d_.*x_]),x_Symbol] :=
  B*x/b + Dist[(b*A-a*B)/b,Int[1/(a+b*sin[c+d*x]),x]] /;
FreeQ[{a,b,c,d,A,B},x] && NonzeroQ[a^2-b^2] && NonzeroQ[b*A-a*B]
```

- **Derivation: Algebraic expansion**
- **Basis: $\frac{A+B/z}{a+b/z} = \frac{A}{a} - \frac{(b A-a B)}{a (b+a z)}$**
- **Rule d3: If $a^2 - b^2 \neq 0 \land b A - a B \neq 0$, then**

$$\int \frac{A + B \operatorname{Csc}[c + d x]}{a + b \operatorname{Csc}[c + d x]} \, dx \longrightarrow \frac{A x}{a} - \frac{b A - a B}{a} \int \frac{1}{b + a \operatorname{Sin}[c + d x]} \, dx$$

- **Program code:**

```
Int[(A_.+B_.*sin[c_.+d_.*x_]^(-1))/(a_+b_.*sin[c_.+d_.*x_]^(-1)),x_Symbol] :=
  A*x/a - Dist[(b*A-a*B)/a,Int[1/(b+a*sin[c+d*x]),x]] /;
FreeQ[{a,b,c,d,A,B},x] && NonzeroQ[a^2-b^2] && NonzeroQ[b*A-a*B]
```

Rule e: $\int \frac{A + B \operatorname{Sin}[c + d x]}{\sqrt{a + b \operatorname{Sin}[c + d x]}} \, dx$

- **Derivation: Algebraic expansion**

- **Basis:** $\frac{A+B z}{\sqrt{a+b z}} = \frac{B}{b} \sqrt{a + b z} + \frac{(b A - a B)}{b} \frac{1}{\sqrt{a+b z}}$

- **Rule e: If** $a^2 - b^2 \neq 0 \land b A - a B \neq 0$**, then**

$$\int \frac{A + B \operatorname{Sin}[c + d x]}{\sqrt{a + b \operatorname{Sin}[c + d x]}} \, dx \longrightarrow \frac{B}{b} \int \sqrt{a + b \operatorname{Sin}[c + d x]} \, dx + \frac{b A - a B}{b} \int \frac{1}{\sqrt{a + b \operatorname{Sin}[c + d x]}} \, dx$$

- **Program code:**

```
Int[(A_.+B_.*sin[c_.+d_.*x_])/Sqrt[a_.+b_.*sin[c_.+d_.*x_]],x_Symbol] :=
  Dist[B/b,Int[Sqrt[a+b*sin[c+d*x]],x]] +
  Dist[(b*A-a*B)/b,Int[1/Sqrt[a+b*sin[c+d*x]],x]] /;
FreeQ[{a,b,c,d,A,B},x] && NonzeroQ[a^2-b^2] && NonzeroQ[b*A-a*B]
```

Rule f: $\int (A + B\,\mathrm{Csc}[c + d\,x])\,(a + b\,\mathrm{Csc}[c + d\,x])^{n/2}\,dx$

- **Derivation: Piecewise constant extraction**
- **Basis:** $\partial_z \frac{1}{\sqrt{f[z]}\,\sqrt{b/f[z]}} = 0$
- **Rule f1:**

$$\int \frac{A + B\,\mathrm{Csc}[c + d\,x]}{\sqrt{b\,\mathrm{Csc}[c + d\,x]}}\,dx \longrightarrow \frac{1}{\sqrt{\mathrm{Sin}[c + d\,x]}\,\sqrt{b\,\mathrm{Csc}[c + d\,x]}}\int \frac{B + A\,\mathrm{Sin}[c + d\,x]}{\sqrt{\mathrm{Sin}[c + d\,x]}}\,dx$$

- **Program code:**

```
Int[(A_+B_.*sin[c_.+d_.*x_]^(-1))/Sqrt[b_.*sin[c_.+d_.*x_]^(-1)],x_Symbol] :=
  Dist[1/(Sqrt[Sin[c+d*x]]*Sqrt[b*Csc[c+d*x]]),Int[(B+A*sin[c+d*x])/Sqrt[sin[c+d*x]],x]] /;
FreeQ[{b,c,d,A,B},x]
```

- **Derivation: Piecewise constant extraction**
- **Basis:** $\partial_z \frac{\sqrt{b+a\,f[z]}}{\sqrt{f[z]}\,\sqrt{a+b/f[z]}} = 0$
- **Rule f2: If** $a^2 - b^2 \neq 0 \bigwedge b\,A - a\,B \neq 0 \bigwedge n - \frac{1}{2} \in \mathbb{Z} \bigwedge -2 < n < 1$**, then**

$$\int (A + B\,\mathrm{Csc}[c + d\,x])\,(a + b\,\mathrm{Csc}[c + d\,x])^n\,dx \longrightarrow$$

$$\frac{\sqrt{b + a\,\mathrm{Sin}[c + d\,x]}}{\sqrt{\mathrm{Sin}[c + d\,x]}\,\sqrt{a + b\,\mathrm{Csc}[c + d\,x]}}\int \frac{(B + A\,\mathrm{Sin}[c + d\,x])\,(b + a\,\mathrm{Sin}[c + d\,x])^n}{\mathrm{Sin}[c + d\,x]^{n+1}}\,dx$$

- **Program code:**

```
Int[(A_.+B_.*sin[c_.+d_.*x_]^(-1))*(a_+b_.*sin[c_.+d_.*x_]^(-1))^n_,x_Symbol] :=
  Dist[Sqrt[b+a*Sin[c+d*x]]/(Sqrt[Sin[c+d*x]]*Sqrt[a+b*Csc[c+d*x]]),
    Int[(B+A*sin[c+d*x])*(b+a*sin[c+d*x])^n/sin[c+d*x]^(n+1),x]] /;
FreeQ[{a,b,c,d,A,B},x] && NonzeroQ[a^2-b^2] && NonzeroQ[b*A-a*B] && IntegerQ[n-1/2] && -2<n<1
```

$$\int \left(A + B\,\mathrm{Sin}[c + d\,x]^k\right)\,\left(a + b\,\mathrm{Sin}[c + d\,x]^k\right)^n\,dx$$

- **Derivation: Algebraic simplification**
- **Basis: If** $b\,A - a\,B = 0$**, then** $(A + B\,z)\,(a + b\,z)^n = \frac{B}{b}\,(a + b\,z)^{n+1}$
- **Rule: If** $k^2 = 1 \wedge b\,A = a\,B \wedge n < 0$**, then**

$$\int \left(A + B\,\mathrm{Sin}[c + d\,x]^k\right)\,\left(a + b\,\mathrm{Sin}[c + d\,x]^k\right)^n\,dx \longrightarrow \frac{B}{b}\int \left(a + b\,\mathrm{Sin}[c + d\,x]^k\right)^{n+1}\,dx$$

- **Program code:**

```
Int[(A_.+B_.*sin[c_.+d_.*x_]^k_.)*(a_.+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  Dist[B/b,Int[(a+b*sin[c+d*x]^k)^(n+1),x]] /;
FreeQ[{a,b,c,d,A,B,n},x] && OneQ[k^2] && ZeroQ[b*A-a*B] && RationalQ[n] && n<0
```

Rules 17 – 18: $\int (A + B\,\mathrm{Csc}[c + d\,x])\,(a + b\,\mathrm{Csc}[c + d\,x])^n\,dx$

- **Derivation: Rule 6 with $m = 0$ and $k = -1$**
- **Rule 17: If $a^2 - b^2 \neq 0 \land b\,A - a\,B \neq 0 \land n < -1$, then**

$$\int (A + B\,\mathrm{Csc}[c + d\,x])\,(a + b\,\mathrm{Csc}[c + d\,x])^n\,dx \longrightarrow$$

$$\frac{b\,(b\,A - a\,B)\,\mathrm{Cot}[c + d\,x]\,(a + b\,\mathrm{Csc}[c + d\,x])^{n+1}}{a\,d\,(n + 1)\,\left(a^2 - b^2\right)} + \frac{1}{a\,(n + 1)\,\left(a^2 - b^2\right)}\,\cdot$$

$$\int \left(A\,\left(a^2 - b^2\right)\,(n + 1) - a\,(b\,A - a\,B)\,(n + 1)\,\mathrm{Csc}[c + d\,x] + b\,(b\,A - a\,B)\,(n + 2)\,\mathrm{Csc}[c + d\,x]^2\right)$$
$$(a + b\,\mathrm{Csc}[c + d\,x])^{n+1}\,dx$$

- **Program code:**

```
Int[(A_+B_.*sin[c_.+d_.*x_]^(-1))*(a_+b_.*sin[c_.+d_.*x_]^(-1))^n_,x_Symbol] :=
  b*(b*A-a*B)*Cot[c+d*x]*(a+b*Csc[c+d*x])^(n+1)/(a*d*(n+1)*(a^2-b^2)) +
  Dist[1/(a*(n+1)*(a^2-b^2)),
    Int[Sim[A*(a^2-b^2)*(n+1)-(a*(b*A-a*B)*(n+1))*sin[c+d*x]^(-1)+
        (b*(b*A-a*B)*(n+2))*sin[c+d*x]^(-2),x]*
      (a+b*sin[c+d*x]^(-1))^(n+1),x]] /;
FreeQ[{a,b,c,d,A,B},x] && NonzeroQ[a^2-b^2] && NonzeroQ[b*A-a*B] && RationalQ[n] && n<-1
```

- **Derivation: Rule 3a with $m = 0$ and $k = -1$**
- **Rule 18: If $a^2 - b^2 \neq 0 \land n > 1$, then**

$$\int (A + B\,\mathrm{Csc}[c + d\,x])\,(a + b\,\mathrm{Csc}[c + d\,x])^n\,dx \longrightarrow$$

$$-\frac{b\,B\,\mathrm{Cot}[c + d\,x]\,(a + b\,\mathrm{Csc}[c + d\,x])^{n-1}}{d\,n} + \frac{1}{n}\,\cdot$$

$$\int \left(a^2\,A\,n + \left(b^2\,B\,(n - 1) + 2\,a\,A\,b\,n + a^2\,B\,n\right)\,\mathrm{Csc}[c + d\,x] + b\,(b\,A\,n + a\,B\,(2\,n - 1))\,\mathrm{Csc}[c + d\,x]^2\right)$$
$$(a + b\,\mathrm{Csc}[c + d\,x])^{n-2}\,dx$$

- **Program code:**

```
Int[(A_+B_.*sin[c_.+d_.*x_]^(-1))*(a_+b_.*sin[c_.+d_.*x_]^(-1))^n_,x_Symbol] :=
  -b*B*Cot[c+d*x]*(a+b*Csc[c+d*x])^(n-1)/(d*n) +
  Dist[1/n,
    Int[Sim[a^2*A*n+(b^2*B*(n-1)+2*a*A*b*n+a^2*B*n)*sin[c+d*x]^(-1)+
        (b*(b*A*n+a*B*(2*n-1)))*sin[c+d*x]^(-2),x]*
      (a+b*sin[c+d*x]^(-1))^(n-2),x]] /;
FreeQ[{a,b,c,d,A,B},x] && NonzeroQ[a^2-b^2] && RationalQ[n] && n>1
```

Rules 15 – 16: $\int \left(A + B\,\text{Sin}[c + d\,x]^k\right) \left(b\,\text{Sin}[c + d\,x]^k\right)^n dx$

- **Derivation: Rule 10a inverted**

- **Rule 15: If $k^2 = 1 \land n < -1$, then**

$$\int \left(A + B\,\text{Sin}[c + d\,x]^k\right) \left(b\,\text{Sin}[c + d\,x]^k\right)^n dx \longrightarrow \frac{2\,A\,\text{Cos}[c + d\,x] \left(b\,\text{Sin}[c + d\,x]^k\right)^{n+1}}{b\,d\,(2\,n + k + 1)} + \frac{1}{b\,(2\,n + k + 1)} \int \left(B\,(2\,n + k + 1) + A\,(2\,n + k + 3)\,\text{Sin}[c + d\,x]^k\right) \left(b\,\text{Sin}[c + d\,x]^k\right)^{n+1} dx$$

- **Program code:**

```
Int[(A_+B_.*sin[c_.+d_.*x_]^k_.)*(b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  2*A*Cos[c+d*x]*(b*Sin[c+d*x]^k)^(n+1)/(b*d*(2*n+k+1)) +
  Dist[1/(b*(2*n+k+1)),
    Int[Sim[B*(2*n+k+1)+A*(2*n+k+3)*sin[c+d*x]^k,x]*(b*sin[c+d*x]^k)^(n+1),x]] /;
FreeQ[{b,c,d,A,B},x] && OneQ[k^2] && RationalQ[n] && n<-1
```

- **Derivation: Rule 3a or 3b with $m = 0$ and $a = 0$**

- **Rule 16: If $k^2 = 1 \land n > 0$, then**

$$\int \left(A + B\,\text{Sin}[c + d\,x]^k\right) \left(b\,\text{Sin}[c + d\,x]^k\right)^n dx \longrightarrow -\frac{2\,B\,\text{Cos}[c + d\,x] \left(b\,\text{Sin}[c + d\,x]^k\right)^{n}}{d\,(2\,n + k + 1)} + \frac{1}{2\,n + k + 1} \int \left(b\,B\,(2\,n + k - 1) + b\,A\,(2\,n + k + 1)\,\text{Sin}[c + d\,x]^k\right) \left(b\,\text{Sin}[c + d\,x]^k\right)^{n-1} dx$$

- **Program code:**

```
Int[(A_+B_.*sin[c_.+d_.*x_]^k_.)*(b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  -2*B*Cos[c+d*x]*(b*Sin[c+d*x]^k)^n/(d*(2*n+k+1)) +
  Dist[1/(2*n+k+1),
    Int[Sim[b*B*(2*n+k-1)+b*A*(2*n+k+1)*sin[c+d*x]^k,x]*(b*sin[c+d*x]^k)^(n-1),x]] /;
FreeQ[{b,c,d,A,B},x] && OneQ[k^2] && RationalQ[n] && n>0
```

Integration Rules for

$$\int\left(\sin^j(z)\right)^m\left(A+B\sin^k(z)\right)\left(a+b\sin^k(z)\right)^n\,dz \text{ when } j^2=1\bigwedge k^2=1$$

Rule g: $\int \frac{A+B\,\mathrm{Sin}[c+d\,x]}{\mathrm{Sin}[c+d\,x]\,(a+b\,\mathrm{Sin}[c+d\,x])}\,dx$

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{A+B\,z}{z\,(a+b\,z)} = \frac{A}{a\,z} - \frac{b\,A-a\,B}{a\,(a+b\,z)}$
- **Rule g: If** $a\,B-b\,A\neq 0$**, then**

$$\int \frac{A+B\,\mathrm{Sin}[c+d\,x]}{\mathrm{Sin}[c+d\,x]\,(a+b\,\mathrm{Sin}[c+d\,x])}\,dx \longrightarrow \frac{A}{a}\int\frac{1}{\mathrm{Sin}[c+d\,x]}\,dx - \frac{b\,A-a\,B}{a}\int\frac{1}{a+b\,\mathrm{Sin}[c+d\,x]}\,dx$$

- **Program code:**

```
Int[(A_+B_.*sin[c_.+d_.*x_])/(sin[c_.+d_.*x_]*(a_+b_.*sin[c_.+d_.*x_])),x_Symbol] :=
  Dist[A/a,Int[1/sin[c+d*x],x]] -
  Dist[(b*A-a*B)/a,Int[1/(a+b*sin[c+d*x]),x]] /;
FreeQ[{a,b,c,d,A,B},x] && NonzeroQ[b*A-a*B]
```

$$\text{Rule h: } \int \frac{\text{Sin}[c+d\,x]^{m/2}\,(A+B\,\text{Sin}[c+d\,x])}{a+b\,\text{Sin}[c+d\,x]}\,dx$$

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{A+B\,z}{a+b\,z} = \frac{B}{b} + \frac{b\,A-a\,B}{b\,(a+b\,z)}$
- **Rule h: If** $a^2-b^2 \neq 0 \wedge b\,A-a\,B \neq 0 \wedge m^2 = 1$**, then**

$$\int \frac{\text{Sin}[c+d\,x]^{m/2}\,(A+B\,\text{Sin}[c+d\,x])}{a+b\,\text{Sin}[c+d\,x]}\,dx \longrightarrow \frac{B}{b}\int \text{Sin}[c+d\,x]^{m/2}\,dx + \frac{b\,A-a\,B}{b}\int \frac{\text{Sin}[c+d\,x]^{m/2}}{a+b\,\text{Sin}[c+d\,x]}\,dx$$

- **Program code:**

```
Int[sin[c_.+d_.*x_]^m_*(A_+B_.*sin[c_.+d_.*x_])/(a_+b_.*sin[c_.+d_.*x_]),x_Symbol] :=
  Dist[B/b,Int[sin[c+d*x]^m,x]] +
  Dist[(b*A-a*B)/b,Int[sin[c+d*x]^m/(a+b*sin[c+d*x]),x]] /;
FreeQ[{a,b,c,d,A,B},x] && NonzeroQ[a^2-b^2] && NonzeroQ[b*A-a*B] && ZeroQ[m^2-1/4]
```

Rule i: $\int \frac{(A + B \operatorname{Sin}[c + d x]) (a + b \operatorname{Sin}[c + d x])^{n/2}}{\operatorname{Sin}[c + d x]} \, dx$

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{A+B z}{z} = B + A \frac{1}{z}$
- **Rule i: If** $a^2 - b^2 \neq 0 \wedge n^2 = 1$, **then**

$$\int \frac{(A + B \operatorname{Sin}[c + d x]) (a + b \operatorname{Sin}[c + d x])^{n/2}}{\operatorname{Sin}[c + d x]} \, dx \longrightarrow$$
$$B \int (a + b \operatorname{Sin}[c + d x])^{n/2} \, dx + A \int \frac{(a + b \operatorname{Sin}[c + d x])^{n/2}}{\operatorname{Sin}[c + d x]} \, dx$$

- **Program code:**

```
Int[(A_+B_.*sin[c_.+d_.*x_])*(a_+b_.*sin[c_.+d_.*x_])^n_/sin[c_.+d_.*x_],x_Symbol] :=
  Dist[B,Int[(a+b*sin[c+d*x])^n,x]] +
  Dist[A,Int[(a+b*sin[c+d*x])^n/sin[c+d*x],x]] /;
FreeQ[{a,b,c,d,A,B},x] && NonzeroQ[a^2-b^2] && ZeroQ[n^2-1/4]
```

$$\text{Rule j: } \int \frac{A + B\,\text{Sin}[c + d\,x]}{\sqrt{\text{Sin}[c + d\,x]}\,\sqrt{a + b\,\text{Sin}[c + d\,x]}}\,dx$$

- **Derivation: Algebraic expansion**
- **Rule j: If $a^2 - b^2 \neq 0 \land A - B \neq 0$, then**

$$\int \frac{A + B\,\text{Sin}[c + d\,x]}{\sqrt{\text{Sin}[c + d\,x]}\,\sqrt{a + b\,\text{Sin}[c + d\,x]}}\,dx \longrightarrow$$
$$B \int \frac{1 + \text{Sin}[c + d\,x]}{\sqrt{\text{Sin}[c + d\,x]}\,\sqrt{a + b\,\text{Sin}[c + d\,x]}}\,dx + (A - B) \int \frac{1}{\sqrt{\text{Sin}[c + d\,x]}\,\sqrt{a + b\,\text{Sin}[c + d\,x]}}\,dx$$

- **Program code:**

```
Int[(A_+B_.*sin[c_.+d_.*x_])/(Sqrt[sin[c_.+d_.*x_]]*Sqrt[a_+b_.*sin[c_.+d_.*x_]]),x_Symbol] :=
  B*Int[(1+sin[c+d*x])/(Sqrt[sin[c+d*x]]*Sqrt[a+b*sin[c+d*x]]),x] +
  (A-B)*Int[1/(Sqrt[sin[c+d*x]]*Sqrt[a+b*sin[c+d*x]]),x] /;
FreeQ[{a,b,c,d,A,B},x] && NonzeroQ[a^2-b^2] && NonzeroQ[A-B]
```

Rule k: $\int \frac{(A+B\,\text{Sin}[c+d\,x])\,\sqrt{a+b\,\text{Sin}[c+d\,x]}}{\sqrt{e+f\,\text{Sin}[c+d\,x]}}\,dx$

- **Derivation: Algebraic transformation**

- **Basis:** $(A+B\,z)\,\sqrt{a+b\,z} = \frac{a\,A+(b\,A+a\,B)\,z+b\,B\,z^2}{\sqrt{a+b\,z}}$

- **Rule k: If** $a^2-b^2\neq 0\,\wedge\,e^2-f^2\neq 0$**, then**

$$\int \frac{(A+B\,\text{Sin}[c+d\,x])\,\sqrt{a+b\,\text{Sin}[c+d\,x]}}{\sqrt{e+f\,\text{Sin}[c+d\,x]}}\,dx \rightarrow \int \frac{a\,A+(b\,A+a\,B)\,\text{Sin}[c+d\,x]+b\,B\,\text{Sin}[c+d\,x]^2}{\sqrt{a+b\,\text{Sin}[c+d\,x]}\,\sqrt{e+f\,\text{Sin}[c+d\,x]}}\,dx$$

- **Program code:**

```
Int[(A_+B_.*sin[c_.+d_.*x_])*Sqrt[a_.+b_.*sin[c_.+d_.*x_]]/Sqrt[e_.+f_.*sin[c_.+d_.*x_]],x_Symbol] :=
  Int[(a*A+(b*A+a*B)*sin[c+d*x]+b*B*sin[c+d*x]^2)/(Sqrt[a+b*sin[c+d*x]]*Sqrt[e+f*sin[c+d*x]]),x] /;
FreeQ[{a,b,c,d,e,f,A,B},x] && NonzeroQ[a^2-b^2] && NonzeroQ[e^2-f^2]
```

Rule l: $\int \frac{A+B \operatorname{Sin}[c+d x]}{\operatorname{Sin}[c+d x]^{3/2} \sqrt{a+b \operatorname{Sin}[c+d x]}} \, dx$

- **Note: This rule is not essential, but produces simpler results.**
- **Rule l1: If $a^2-b^2 \neq 0$, then**

$$\int \frac{A-A \operatorname{Sin}[c+d x]}{\operatorname{Sin}[c+d x]^{3/2} \sqrt{a+b \operatorname{Sin}[c+d x]}} \, dx \longrightarrow$$

$$\frac{2 A \sqrt{a+b \operatorname{Sin}[c+d x]} \operatorname{Tan}\left[\frac{1}{2}\left(c-\frac{\pi}{2}+d x\right)\right]}{a d \sqrt{\operatorname{Sin}[c+d x]}}-\frac{2 A}{a} \int \frac{\sqrt{a+b \operatorname{Sin}[c+d x]}}{\sqrt{\operatorname{Sin}[c+d x]}\,(1+\operatorname{Sin}[c+d x])} \, dx$$

- **Program code:**

```
Int[(A_+B_.*sin[c_.+d_.*x_])/(sin[c_.+d_.*x_]^(3/2)*Sqrt[a_+b_.*sin[c_.+d_.*x_]]),x_Symbol] :=
  2*A*Sqrt[a+b*Sin[c+d*x]]*Tan[(c-Pi/2+d*x)/2]/(a*d*Sqrt[Sin[c+d*x]]) -
  2*A/a*Int[Sqrt[a+b*sin[c+d*x]]/(Sqrt[sin[c+d*x]]*(1+sin[c+d*x])),x] /;
FreeQ[{a,b,c,d,A,B},x] && NonzeroQ[a^2-b^2] && ZeroQ[A+B]
```

- **Derivation: Algebraic expansion**
- **Note: This rule is not essential, but produces simpler results.**
- **Rule l2: If $a^2-b^2 \neq 0 \land A+B \neq 0$, then**

$$\int \frac{A+B \operatorname{Sin}[c+d x]}{\operatorname{Sin}[c+d x]^{3/2} \sqrt{a+b \operatorname{Sin}[c+d x]}} \, dx \longrightarrow$$

$$(A+B) \int \frac{1}{\sqrt{\operatorname{Sin}[c+d x]} \sqrt{a+b \operatorname{Sin}[c+d x]}} \, dx+A \int \frac{1-\operatorname{Sin}[c+d x]}{\operatorname{Sin}[c+d x]^{3/2} \sqrt{a+b \operatorname{Sin}[c+d x]}} \, dx$$

- **Program code:**

```
Int[(A_+B_.*sin[c_.+d_.*x_])/(sin[c_.+d_.*x_]^(3/2)*Sqrt[a_+b_.*sin[c_.+d_.*x_]]),x_Symbol] :=
  Dist[A+B,Int[1/(Sqrt[sin[c+d*x]]*Sqrt[a+b*sin[c+d*x]]),x]] +
  Dist[A,Int[(1-sin[c+d*x])/(sin[c+d*x]^(3/2)*Sqrt[a+b*sin[c+d*x]]),x]] /;
FreeQ[{a,b,c,d,A,B},x] && NonzeroQ[a^2-b^2] && NonzeroQ[A+B]
```

Rule m: $\int \frac{A + B \sin[c + d x]}{\sqrt{\sin[c + d x]} \, (a + b \sin[c + d x])^{3/2}} \, dx$

- **Note: This rule is not essential, but produces simpler results.**
- **Rule m1: If $a^2 - b^2 \neq 0$, then**

$$\int \frac{A + A \sin[c + d x]}{\sqrt{\sin[c + d x]} \, (a + b \sin[c + d x])^{3/2}} \, dx \longrightarrow$$

$$\frac{2 A (a - b) \sqrt{\sin[c + d x]} \, \tan\left[\frac{1}{2} \left(c - \frac{\pi}{2} + d x\right)\right]}{a d (a + b) \sqrt{a + b \sin[c + d x]}} + \frac{2 A}{a (a + b)} \int \frac{\sqrt{a + b \sin[c + d x]}}{\sqrt{\sin[c + d x]} \, (1 + \sin[c + d x])} \, dx$$

- **Program code:**

```
Int[(A_+B_.*sin[c_.+d_.*x_])/(Sqrt[sin[c_.+d_.*x_]]*(a_+b_.*sin[c_.+d_.*x_])^(3/2)),x_Symbol] :=
  2*A*(a-b)*Sqrt[Sin[c+d*x]]*Tan[(c-Pi/2+d*x)/2]/(a*d*(a+b)*Sqrt[a+b*Sin[c+d*x]]) +
  Dist[2*A/(a*(a+b)),Int[Sqrt[a+b*sin[c+d*x]]/(Sqrt[sin[c+d*x]]*(1+sin[c+d*x])),x]] /;
FreeQ[{a,b,c,d,A,B},x] && NonzeroQ[a^2-b^2] && ZeroQ[A-B]
```

- **Derivation: Algebraic expansion**
- **Note: This rule is not essential, but produces simpler results.**
- **Rule m2: If $a^2 - b^2 \neq 0 \wedge b A - a B \neq 0 \wedge A - B \neq 0$, then**

$$\int \frac{A + B \sin[c + d x]}{\sqrt{\sin[c + d x]} \, (a + b \sin[c + d x])^{3/2}} \, dx \longrightarrow$$

$$\frac{A - B}{a - b} \int \frac{1}{\sqrt{\sin[c + d x]} \, \sqrt{a + b \sin[c + d x]}} \, dx - \frac{b A - a B}{a - b} \int \frac{1 + \sin[c + d x]}{\sqrt{\sin[c + d x]} \, (a + b \sin[c + d x])^{3/2}} \, dx$$

- **Program code:**

```
Int[(A_+B_.*sin[c_.+d_.*x_])/(Sqrt[sin[c_.+d_.*x_]]*(a_+b_.*sin[c_.+d_.*x_])^(3/2)),x_Symbol] :=
  Dist[(A-B)/(a-b),Int[1/(Sqrt[sin[c+d*x]]*Sqrt[a+b*sin[c+d*x]]),x]] -
  Dist[(b*A-a*B)/(a-b),Int[(1+sin[c+d*x])/(Sqrt[sin[c+d*x]]*(a+b*sin[c+d*x])^(3/2)),x]] /;
FreeQ[{a,b,c,d,A,B},x] && NonzeroQ[a^2-b^2] && NonzeroQ[b*A-a*B] && NonzeroQ[A-B]
```

$$\text{Rule n: } \int \frac{(A + B\,\text{Sin}[c + d\,x])\,\sqrt{a + b\,\text{Sin}[c + d\,x]}}{\text{Sin}[c + d\,x]^{3/2}}\,dx$$

- **Derivation: Algebraic expansion**
- **Note: This rule is not essential, but produces simpler results.**
- **Rule n: If $a^2 - b^2 \neq 0$, then**

$$\int \frac{(A + B\,\text{Sin}[c + d\,x])\,\sqrt{a + b\,\text{Sin}[c + d\,x]}}{\text{Sin}[c + d\,x]^{3/2}}\,dx \longrightarrow$$

$$(b\,(A - B) + a\,(A + B))\int \frac{1}{\sqrt{\text{Sin}[c + d\,x]}\,\sqrt{a + b\,\text{Sin}[c + d\,x]}}\,dx +$$

$$\int \frac{a\,A - (a\,A - b\,B)\,\text{Sin}[c + d\,x] + b\,B\,\text{Sin}[c + d\,x]^2}{\text{Sin}[c + d\,x]^{3/2}\,\sqrt{a + b\,\text{Sin}[c + d\,x]}}\,dx$$

- **Program code:**

```
Int[(A_+B_.*sin[c_.+d_.*x_])*Sqrt[a_+b_.*sin[c_.+d_.*x_]]/sin[c_.+d_.*x_]^(3/2),x_Symbol] :=
  (b*(A-B)+a*(A+B))*Int[1/(Sqrt[sin[c+d*x]]*Sqrt[a+b*sin[c+d*x]]),x] +
  Int[Sim[a*A-(a*A-b*B)*sin[c+d*x]+b*B*sin[c+d*x]^2,x]/(sin[c+d*x]^(3/2)*Sqrt[a+b*sin[c+d*x]]),x] /;
FreeQ[{a,b,c,d,A,B},x] && NonzeroQ[a^2-b^2]
```

Rule o: $\int \frac{\sqrt{\text{Sin}[c+d\,x]}\,(A+B\,\text{Sin}[c+d\,x])}{(a+b\,\text{Sin}[c+d\,x])^{3/2}}\,dx$

- **Derivation: Algebraic expansion**
- **Note: This rule is not essential, but produces simpler results.**
- **Rule o: If** $a^2-b^2\neq 0\ \wedge\ b\,A-a\,B\neq 0$**, then**

$$\int \frac{\sqrt{\text{Sin}[c+d\,x]}\,(A+B\,\text{Sin}[c+d\,x])}{(a+b\,\text{Sin}[c+d\,x])^{3/2}}\,dx\ \longrightarrow$$
$$\frac{B}{b}\int \frac{1+\text{Sin}[c+d\,x]}{\sqrt{\text{Sin}[c+d\,x]}\,\sqrt{a+b\,\text{Sin}[c+d\,x]}}\,dx+\frac{1}{b}\int \frac{-a\,B+(A\,b-(a+b)\,B)\,\text{Sin}[c+d\,x]}{\sqrt{\text{Sin}[c+d\,x]}\,(a+b\,\text{Sin}[c+d\,x])^{3/2}}\,dx$$

- **Program code:**

```
Int[Sqrt[sin[c_.+d_.*x_]]*(A_+B_.*sin[c_.+d_.*x_])/(a_+b_.*sin[c_.+d_.*x_])^(3/2),x_Symbol] :=
  B/b*Int[(1+sin[c+d*x])/(Sqrt[sin[c+d*x]]*Sqrt[a+b*sin[c+d*x]]),x] +
  Dist[1/b,Int[Sim[-a*B+(A*b-(a+b)*B)*sin[c+d*x],x]/(Sqrt[sin[c+d*x]]*(a+b*sin[c+d*x])^(3/2)),x]] /;
FreeQ[{a,b,c,d,A,B},x] && NonzeroQ[a^2-b^2] && NonzeroQ[b*A-a*B]
```

Rule p: $\int \left(\mathrm{Sin}[c+d\,x]^j\right)^m \left(A+B\,\mathrm{Sin}[c+d\,x]^k\right) \left(b\,\mathrm{Sin}[c+d\,x]^k\right)^n \, dx$

- **Derivation: Algebraic simplification**
- **Rule p1: If** $k^2 = 1 \land m \in \mathbb{Z}$, **then**

$$\int \mathrm{Sin}[c+d\,x]^m \left(A+B\,\mathrm{Sin}[c+d\,x]^k\right) \left(b\,\mathrm{Sin}[c+d\,x]^k\right)^n dx \longrightarrow \frac{1}{b^{k\,m}} \int \left(A+B\,\mathrm{Sin}[c+d\,x]^k\right) \left(b\,\mathrm{Sin}[c+d\,x]^k\right)^{k\,m+n} dx$$

- **Program code:**

```
Int[sin[c_.+d_.*x_]^m_.*(A_+B_.*sin[c_.+d_.*x_]^k_.)*(b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  Dist[1/b^(k*m),Int[(A+B*sin[c+d*x]^k)*(b*sin[c+d*x]^k)^(k*m+n),x]] /;
FreeQ[{b,c,d,A,B,n},x] && OneQ[k^2] && IntegerQ[m]
```

- **Derivation: Piecewise constant extraction**
- **Basis: If** $j^2 = 1$, **then** $\partial_z \frac{\sqrt{b\,f[z]^k}}{\left(\sqrt{f[z]^j}\right)^{j\,k}} = 0$
- **Rule p2: If** $j^2 = k^2 = 1 \bigwedge m - \frac{1}{2} \in \mathbb{Z} \bigwedge n - \frac{1}{2} \in \mathbb{Z} \bigwedge n > 0$, **then**

$$\int \left(\mathrm{Sin}[c+d\,x]^j\right)^m \left(A+B\,\mathrm{Sin}[c+d\,x]^k\right) \left(b\,\mathrm{Sin}[c+d\,x]^k\right)^n dx \longrightarrow \frac{b^{n-\frac{1}{2}} \sqrt{b\,\mathrm{Sin}[c+d\,x]^k}}{\left(\sqrt{\mathrm{Sin}[c+d\,x]^j}\right)^{j\,k}} \int \mathrm{Sin}[c+d\,x]^{j\,m+k\,n} \left(A+B\,\mathrm{Sin}[c+d\,x]^k\right) dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_*(A_+B_.*sin[c_.+d_.*x_]^k_.)*(b_*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  Dist[b^(n-1/2)*Sqrt[b*Sin[c+d*x]^k]/(Sqrt[Sin[c+d*x]^j])^(j*k),
    Int[sin[c+d*x]^(j*m+k*n)*(A+B*sin[c+d*x]^k),x]] /;
FreeQ[{b,c,d,A,B},x] && OneQ[j^2,k^2] && IntegerQ[m-1/2] && IntegerQ[n-1/2] && n>0
```

- **Derivation: Piecewise constant extraction**

- **Basis: If** $j^2 = 1$**, then** $\partial_z \frac{\left(\sqrt{f[z]^j}\right)^{j k}}{\sqrt{b f[z]^k}} = 0$

- **Rule p3: If** $j^2 = k^2 = 1 \bigwedge m - \frac{1}{2} \in \mathbb{Z} \bigwedge n - \frac{1}{2} \in \mathbb{Z} \bigwedge n < 0$**, then**

$$\int \left(\text{Sin}[c + d x]^j\right)^m \left(A + B \,\text{Sin}[c + d x]^k\right) \left(b \,\text{Sin}[c + d x]^k\right)^n \, dx \longrightarrow$$

$$\frac{b^{n+\frac{1}{2}} \left(\sqrt{\text{Sin}[c + d x]^j}\right)^{j k}}{\sqrt{b \,\text{Sin}[c + d x]^k}} \int \text{Sin}[c + d x]^{j m + k n} \left(A + B \,\text{Sin}[c + d x]^k\right) dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_*(A_+B_.*sin[c_.+d_.*x_]^k_.)*(b_*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  Dist[b^(n+1/2)*(Sqrt[Sin[c+d*x]^j])^(j*k)/Sqrt[b*Sin[c+d*x]^k],
    Int[sin[c+d*x]^(j*m+k*n)*(A+B*sin[c+d*x]^k),x]] /;
FreeQ[{b,c,d,A,B},x] && OneQ[j^2,k^2] && IntegerQ[m-1/2] && IntegerQ[n-1/2] && n<0
```

Rule q: $\int \left(\text{Sin}[c+d\,x]^j\right)^m (A+B\,\text{Csc}[c+d\,x])\,(a+b\,\text{Csc}[c+d\,x])^n\,dx$

- **Derivation: Algebraic simplification**
- **Rule q1: If** $j^2=1 \land a^2-b^2\neq 0 \land -1<m\leq 1$, **then**

$$\int \frac{\left(\text{Sin}[c+d\,x]^j\right)^m (A+B\,\text{Csc}[c+d\,x])}{a+b\,\text{Csc}[c+d\,x]}\,dx \longrightarrow \int \frac{\left(\text{Sin}[c+d\,x]^j\right)^m (B+A\,\text{Sin}[c+d\,x])}{b+a\,\text{Sin}[c+d\,x]}\,dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(A_.+B_.*sin[c_.+d_.*x_]^(-1))/(a_+b_.*sin[c_.+d_.*x_]^(-1)),x_Symbol] :=
  Int[(sin[c+d*x]^j)^m*(B+A*sin[c+d*x])/(b+a*sin[c+d*x]),x] /;
FreeQ[{a,b,c,d,A,B},x] && OneQ[j^2] && NonzeroQ[a^2-b^2] && RationalQ[m] && -1<m≤1
```

- **Derivation: Piecewise constant extraction**
- **Basis:** $\partial_z \frac{\sqrt{b+a\,f[z]}}{\sqrt{f[z]}\,\sqrt{a+b/f[z]}}=0$
- **Rule q2: If** $a^2-b^2\neq 0 \bigwedge m\in\mathbb{Z} \bigwedge n-\frac{1}{2}\in\mathbb{Z} \bigwedge ((m=1 \land -1<n<1) \lor (m=-1 \land -2<n<0))$, **then**

$$\int \text{Sin}[c+d\,x]^m (A+B\,\text{Csc}[c+d\,x])\,(a+b\,\text{Csc}[c+d\,x])^n\,dx \longrightarrow$$

$$\frac{\sqrt{b+a\,\text{Sin}[c+d\,x]}}{\sqrt{\text{Sin}[c+d\,x]}\,\sqrt{a+b\,\text{Csc}[c+d\,x]}}\int \text{Sin}[c+d\,x]^{m-n-1} (B+A\,\text{Sin}[c+d\,x])\,(b+a\,\text{Sin}[c+d\,x])^n\,dx$$

- **Program code:**

```
Int[sin[c_.+d_.*x_]^m_.*(A_.+B_.*sin[c_.+d_.*x_]^(-1))*(a_.+b_.*sin[c_.+d_.*x_]^(-1))^n_,x_Symbol] :=
  Dist[Sqrt[b+a*Sin[c+d*x]]/(Sqrt[Sin[c+d*x]]*Sqrt[a+b*Csc[c+d*x]]),
    Int[sin[c+d*x]^(m-n-1)*(B+A*sin[c+d*x])*(b+a*sin[c+d*x])^n,x]] /;
FreeQ[{a,b,c,d,A,B},x] && NonzeroQ[a^2-b^2] && IntegerQ[m] && IntegerQ[n-1/2] &&
  (m==1 && -1<n<1 || m==-1 && -2<n<0)
```

- **Derivation: Piecewise constant extraction**

- **Basis: If** $j^2 = 1$**, then** $\partial_z \frac{\sqrt{b+a\,f[z]}}{\left(\sqrt{f[z]^j}\right)^j \sqrt{a+b\,f[z]^{-1}}} = 0$

- **Rule q3: If** $j^2 = 1 \bigwedge a^2 - b^2 \neq 0 \bigwedge m - \frac{1}{2} \in \mathbb{Z} \bigwedge n - \frac{1}{2} \in \mathbb{Z} \bigwedge 0 \leq j\,m - n \leq 1$**, then**

$$\int \left(\text{Sin}[c+d\,x]^j\right)^m (A + B\,\text{Csc}[c+d\,x])\,(a + b\,\text{Csc}[c+d\,x])^n\,dx \longrightarrow$$

$$\frac{\sqrt{b + a\,\text{Sin}[c+d\,x]}}{\sqrt{\text{Sin}[c+d\,x]^j}^{\,j}\,\sqrt{a + b\,\text{Csc}[c+d\,x]}} \int \text{Sin}[c+d\,x]^{j\,m-n-1}\,(B + A\,\text{Sin}[c+d\,x])\,(b + a\,\text{Sin}[c+d\,x])^n\,dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_*(A_.+B_.*sin[c_.+d_.*x_]^(-1))*(a_.+b_.*sin[c_.+d_.*x_]^(-1))^n_,x_Symb
  Dist[Sqrt[b+a*Sin[c+d*x]]/((Sqrt[Sin[c+d*x]^j])^j*Sqrt[a+b*Csc[c+d*x]]),
    Int[sin[c+d*x]^(j*m-n-1)*(B+A*sin[c+d*x])*(b+a*sin[c+d*x])^n,x]] /;
FreeQ[{a,b,c,d,A,B},x] && OneQ[j^2] && NonzeroQ[a^2-b^2] &&
  IntegerQ[m-1/2] && IntegerQ[n-1/2] && -1<n<1 && 0≤j*m-n≤1
```

Rule r: $\int \mathrm{Csc}[c+d\,x]^m\,(A+B\,\mathrm{Sin}[c+d\,x])\,(a+b\,\mathrm{Sin}[c+d\,x])^n\,dx$

- **Derivation: Piecewise constant extraction**
- **Basis:** $\partial_z\left(\sqrt{f[z]}\,\sqrt{1/f[z]}\right)=0$
- **Rule r: If** $m-\frac{1}{2}\in\mathbb{Z}\ \bigwedge\ -1<m<2\ \bigwedge\ -2<n<1$**, then**

$$\int \mathrm{Csc}[c+d\,x]^m\,(A+B\,\mathrm{Sin}[c+d\,x])\,(a+b\,\mathrm{Sin}[c+d\,x])^n\,dx\ \longrightarrow$$
$$\sqrt{\mathrm{Csc}[c+d\,x]}\,\sqrt{\mathrm{Sin}[c+d\,x]}\int\frac{(A+B\,\mathrm{Sin}[c+d\,x])\,(a+b\,\mathrm{Sin}[c+d\,x])^n}{\mathrm{Sin}[c+d\,x]^m}\,dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^(-1))^m_*(A_.+B_.*sin[c_.+d_.*x_])*(a_+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  Dist[Sqrt[Csc[c+d*x]]*Sqrt[Sin[c+d*x]],
    Int[(A+B*sin[c+d*x])*(a+b*sin[c+d*x])^n/sin[c+d*x]^m,x]] /;
FreeQ[{a,b,c,d,A,B},x] && IntegerQ[m-1/2] && RationalQ[n] && -1<m<2 && -2<n<1
```

Rules 13 – 14: $\int \text{Sin}[c+d\,x]\,(A+B\,\text{Sin}[c+d\,x])\,(a+b\,\text{Sin}[c+d\,x])^n\,dx$

- **Derivation: Rule 1a with $j\,m=1$ and $k=1$**
- **Rule 13: If $a^2-b^2\neq 0\,\land\,b\,A-a\,B\neq 0\,\land\,n<-1$, then**

$$\int \text{Sin}[c+d\,x]\,(A+B\,\text{Sin}[c+d\,x])\,(a+b\,\text{Sin}[c+d\,x])^n\,dx\,\longrightarrow$$

$$\frac{a\,(b\,A-a\,B)\,\text{Cos}[c+d\,x]\,(a+b\,\text{Sin}[c+d\,x])^{n+1}}{b\,d\,(n+1)\,\left(a^2-b^2\right)}-\frac{1}{b\,(n+1)\,\left(a^2-b^2\right)}\,\cdot$$

$$\int\left(b\,(n+1)\,(b\,A-a\,B)+\left(a^2\,B-a\,b\,A\,(n+2)+b^2\,B\,(n+1)\right)\text{Sin}[c+d\,x]\right)\,(a+b\,\text{Sin}[c+d\,x])^{n+1}\,dx$$

- **Program code:**

```
Int[sin[c_.+d_.*x_]*(A_+B_.*sin[c_.+d_.*x_])*(a_+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  a*(b*A-a*B)*Cos[c+d*x]*(a+b*Sin[c+d*x])^(n+1)/(b*d*(n+1)*(a^2-b^2)) -
  Dist[1/(b*(n+1)*(a^2-b^2)),
    Int[Sim[b*(n+1)*(b*A-a*B)+(a^2*B-a*b*A*(n+2)+b^2*B*(n+1))*sin[c+d*x],x]*
      (a+b*sin[c+d*x])^(n+1),x]] /;
FreeQ[{a,b,c,d,A,B},x] && NonzeroQ[a^2-b^2] && NonzeroQ[b*A-a*B] && RationalQ[n] && n<-1
```

- **Derivation: Rule 14b with $n=-1$**
- **Note: This is an unnecessary special case of rule 14b, but it saves a trivial step.**
- **Rule 14a: If $b\,A-a\,B\neq 0$, then**

$$\int\frac{\text{Sin}[c+d\,x]\,(A+B\,\text{Sin}[c+d\,x])}{a+b\,\text{Sin}[c+d\,x]}\,dx\,\longrightarrow\,-\frac{B\,\text{Cos}[c+d\,x]}{b\,d}+\frac{b\,A-a\,B}{b}\int\frac{\text{Sin}[c+d\,x]}{a+b\,\text{Sin}[c+d\,x]}\,dx$$

- **Program code:**

```
Int[sin[c_.+d_.*x_]*(A_+B_.*sin[c_.+d_.*x_])/(a_+b_.*sin[c_.+d_.*x_]),x_Symbol] :=
  -B*Cos[c+d*x]/(b*d) +
  Dist[(b*A-a*B)/b,Int[sin[c+d*x]/(a+b*sin[c+d*x]),x]] /;
FreeQ[{a,b,c,d,A,B},x] && NonzeroQ[b*A-a*B]
```

- **Derivation: Rule 2 with $j\,m=1$ and $k=1$**
- **Rule 14b: If $n>-1\,\land\,n\neq 1$, then**

$$\int \mathrm{Sin}[c+d\,x]\,(A+B\,\mathrm{Sin}[c+d\,x])\,(a+b\,\mathrm{Sin}[c+d\,x])^n\,dx \longrightarrow$$

$$-\frac{B\,\mathrm{Cos}[c+d\,x]\,(a+b\,\mathrm{Sin}[c+d\,x])^{n+1}}{b\,d\,(n+2)}+$$

$$\frac{1}{b\,(n+2)}\int (b\,B\,(n+1)-(a\,B-b\,A\,(n+2))\,\mathrm{Sin}[c+d\,x])\,(a+b\,\mathrm{Sin}[c+d\,x])^n\,dx$$

- **Program code:**

```
Int[sin[c_.+d_.*x_]*(A_+B_.*sin[c_.+d_.*x_])*(a_+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  -B*Cos[c+d*x]*(a+b*Sin[c+d*x])^(n+1)/(b*d*(n+2)) +
  Dist[1/(b*(n+2)),Int[Sim[b*B*(n+1)-(a*B-b*A*(n+2))*sin[c+d*x],x]*(a+b*sin[c+d*x])^n,x]] /;
FreeQ[{a,b,c,d,A,B},x] && RationalQ[n] && n>-1 && n≠1
```

Rules 11 – 12 :

$$\int (\text{Sin}[c+d\,x]^j)^m\,(A+B\,\text{Sin}[c+d\,x]^k)\,(a+b\,\text{Sin}[c+d\,x]^k)\,dx\quad j\,k\,m<-1\,??\,?$$

- **Derivation: Rule 4a with** $n = 1$

- **Rule 11: If** $j^2 = k^2 = 1 \bigwedge a^2 - b^2 \neq 0 \bigwedge b\,A - a\,B \neq 0 \bigwedge j\,k\,m + \frac{k+1}{2} \neq 0 \bigwedge j\,k\,m \leq -1$**, then**

$$\int (\text{Sin}[c+d\,x]^j)^m\,(A+B\,\text{Sin}[c+d\,x]^k)\,(a+b\,\text{Sin}[c+d\,x]^k)\,dx \longrightarrow$$

$$\frac{a\,A\,\text{Cos}[c+d\,x]\,(\text{Sin}[c+d\,x]^j)^{m+j\,k}}{d\left(j\,k\,m+\frac{k+1}{2}\right)}+\frac{1}{j\,k\,m+\frac{k+1}{2}}\cdot$$
$$\int (\text{Sin}[c+d\,x]^j)^{m+j\,k}\left((b\,A+a\,B)\left(j\,k\,m+\frac{k+1}{2}\right)+\left(a\,A\left(j\,k\,m+\frac{k+3}{2}\right)+b\,B\left(j\,k\,m+\frac{k+1}{2}\right)\right)\text{Sin}[c+d\,x]^k\right)dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(A_+B_.*sin[c_.+d_.*x_]^k_.)*(a_+b_.*sin[c_.+d_.*x_]^k_.),x_Symbol] :=
  a*A*Cos[c+d*x]*(Sin[c+d*x]^j)^(m+j*k)/(d*(j*k*m+(k+1)/2)) +
  Dist[1/(j*k*m+(k+1)/2),
    Int[(sin[c+d*x]^j)^(m+j*k)*
      Sim[(b*A+a*B)*(j*k*m+(k+1)/2)+(a*A*(j*k*m+(k+3)/2)+b*B*(j*k*m+(k+1)/2))*sin[c+d*x]^k,x],x]] /;
FreeQ[{a,b,c,d,A,B},x] && OneQ[j^2,k^2] && NonzeroQ[a^2-b^2] &&
  RationalQ[m] && j*k*m+(k+1)/2≠0 && j*k*m≤-1
```

- **Derivation: Rule 3a with** $n = 1$

- **Rule 12: If** $j^2 = k^2 = 1 \wedge a^2 - b^2 \neq 0 \wedge b\,A - a\,B \neq 0 \wedge j\,k\,m \geq -1$**, then**

$$\int (\text{Sin}[c+d\,x]^j)^m\,(A+B\,\text{Sin}[c+d\,x]^k)\,(a+b\,\text{Sin}[c+d\,x]^k)\,dx \longrightarrow$$

$$-\frac{b\,B\,\text{Cos}[c+d\,x]\,(\text{Sin}[c+d\,x]^j)^{m+j\,k}}{d\left(j\,k\,m+\frac{k+3}{2}\right)}+\frac{1}{j\,k\,m+\frac{k+3}{2}}\cdot$$
$$\int (\text{Sin}[c+d\,x]^j)^m\left(a\,A\left(j\,k\,m+\frac{k+3}{2}\right)+b\,B\left(j\,k\,m+\frac{k+1}{2}\right)+(b\,A+a\,B)\left(j\,k\,m+\frac{k+3}{2}\right)\text{Sin}[c+d\,x]^k\right)dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(A_+B_.*sin[c_.+d_.*x_]^k_.)*(a_+b_.*sin[c_.+d_.*x_]^k_.),x_Symbol] :=
  -b*B*Cos[c+d*x]*(Sin[c+d*x]^j)^(m+j*k)/(d*(j*k*m+(k+3)/2)) +
  Dist[1/(j*k*m+(k+3)/2),
    Int[(sin[c+d*x]^j)^m*
      Sim[a*A*(j*k*m+(k+3)/2)+b*B*(j*k*m+(k+1)/2)+(b*A+a*B)*(j*k*m+(k+3)/2)*sin[c+d*x]^k,x],x]] /;
FreeQ[{a,b,c,d,A,B},x] && OneQ[j^2,k^2] && NonzeroQ[a^2-b^2] &&
  RationalQ[m] && j*k*m≥-1
```

Rules 9 – 10: $\int \text{Sin}[c+d\,x]^{\frac{k-1}{2}} \left(A+B\,\text{Sin}[c+d\,x]^k\right) \left(a+b\,\text{Sin}[c+d\,x]^k\right)^n dx$

- **Reference: G&R 2.551.1**
- **Derivation: Rule 1b with** $j\,m=\frac{k-1}{2}$
- **Rule 9: If** $k^2=1\ \land\ a^2-b^2\neq 0\ \land\ b\,A-a\,B\neq 0\ \land\ n<-1$, **then**

$$\int \text{Sin}[c+d\,x]^{\frac{k-1}{2}} \left(A+B\,\text{Sin}[c+d\,x]^k\right) \left(a+b\,\text{Sin}[c+d\,x]^k\right)^n dx \longrightarrow$$

$$-\frac{(b\,A-a\,B)\,\text{Cos}[c+d\,x]\,\text{Sin}[c+d\,x]^{\frac{k-1}{2}} \left(a+b\,\text{Sin}[c+d\,x]^k\right)^{n+1}}{d\,(n+1)\left(a^2-b^2\right)}+\frac{1}{(n+1)\left(a^2-b^2\right)}$$

$$\int \text{Sin}[c+d\,x]^{\frac{k-1}{2}} \left((a\,A-b\,B)\,(n+1)-(b\,A-a\,B)\,(n+2)\,\text{Sin}[c+d\,x]^k\right) \left(a+b\,\text{Sin}[c+d\,x]^k\right)^{n+1} dx$$

- **Program code:**

```
Int[(A_.+B_.*sin[c_.+d_.*x_])*(a_+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  -(b*A-a*B)*Cos[c+d*x]*(a+b*Sin[c+d*x])^(n+1)/(d*(n+1)*(a^2-b^2)) +
  Dist[1/((n+1)*(a^2-b^2)),
    Int[Sim[(a*A-b*B)*(n+1)-(b*A-a*B)*(n+2)*sin[c+d*x],x]*(a+b*sin[c+d*x])^(n+1),x]] /;
FreeQ[{a,b,c,d,A,B},x] && NonzeroQ[a^2-b^2] && NonzeroQ[b*A-a*B] && RationalQ[n] && n<-1
```

```
Int[sin[c_.+d_.*x_]^(-1)*(A_.+B_.*sin[c_.+d_.*x_]^(-1))*(a_+b_.*sin[c_.+d_.*x_]^(-1))^n_,x_Symbol] :
  -(b*A-a*B)*Cot[c+d*x]*(a+b*Csc[c+d*x])^(n+1)/(d*(n+1)*(a^2-b^2)) +
  Dist[1/((n+1)*(a^2-b^2)),
    Int[sin[c+d*x]^(-1)*
      Sim[(a*A-b*B)*(n+1)-(b*A-a*B)*(n+2)*sin[c+d*x]^(-1),x]*(a+b*sin[c+d*x]^(-1))^(n+1),x]] /;
FreeQ[{a,b,c,d,A,B},x] && NonzeroQ[a^2-b^2] && NonzeroQ[b*A-a*B] && RationalQ[n] && n<-1
```

- **Reference: G&R 2.551.1 inverted**
- **Derivation: Rule 3b with** $j\,m = \frac{k-1}{2}$
- **Rule 10: If** $k^2 = 1 \wedge a^2 - b^2 \neq 0 \wedge n > 0 \wedge n \neq 1$**, then**

$$\int \text{Sin}[c+d\,x]^{\frac{k-1}{2}} \left(A+B\,\text{Sin}[c+d\,x]^k\right) \left(a+b\,\text{Sin}[c+d\,x]^k\right)^n dx \longrightarrow$$

$$-\frac{B\,\text{Cos}[c+d\,x]\,\text{Sin}[c+d\,x]^{\frac{k-1}{2}} \left(a+b\,\text{Sin}[c+d\,x]^k\right)^n}{d\,(n+1)} +$$

$$\frac{1}{n+1} \int \text{Sin}[c+d\,x]^{\frac{k-1}{2}} \left(b\,B\,n+a\,A\,(n+1)+(a\,B\,n+b\,A\,(n+1))\,\text{Sin}[c+d\,x]^k\right) \left(a+b\,\text{Sin}[c+d\,x]^k\right)^{n-1} dx$$

- **Program code:**

```
Int[(A_.+B_.*sin[c_.+d_.*x_])*(a_.+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  -B*Cos[c+d*x]*(a+b*Sin[c+d*x])^n/(d*(n+1)) +
  Dist[1/(n+1),
    Int[Sim[b*B*n+a*A*(n+1)+(a*B*n+b*A*(n+1))*sin[c+d*x],x]*(a+b*sin[c+d*x])^(n-1),x]] /;
FreeQ[{a,b,c,d,A,B},x] && RationalQ[n] && n>0 && n≠1
```

```
Int[sin[c_.+d_.*x_]^(-1)*(A_.+B_.*sin[c_.+d_.*x_]^(-1))*(a_+b_.*sin[c_.+d_.*x_]^(-1))^n_,x_Symbol] :
  -B*Cot[c+d*x]*(a+b*Csc[c+d*x])^n/(d*(n+1)) +
  Dist[1/(n+1),
    Int[sin[c+d*x]^(-1)*
      Sim[b*B*n+a*A*(n+1)+(a*B*n+b*A*(n+1))*sin[c+d*x]^(-1),x]*(a+b*sin[c+d*x]^(-1))^(n-1),x]] /;
FreeQ[{a,b,c,d,A,B},x] && RationalQ[n] && n>0 && n≠1
```

Rules 1 – 6: $\int (\text{Sin}[c+d\,x]^j)^m\,(A+B\,\text{Sin}[c+d\,x]^k)\,(a+b\,\text{Sin}[c+d\,x]^k)^n\,dx$

- **Derivation: Rule 1a, 1b or 6 with** $b\,A-a\,B=0$
- **Derivation: Algebraic simplification**
- **Basis: If** $b\,A-a\,B=0$**, then** $A+B\,z=\frac{B}{b}\,(a+b\,z)$
- **Rule: If** $j^2=k^2=1\,\wedge\,b\,A-a\,B=0\,\wedge\,n\le-1$**, then**

$$\int (\text{Sin}[c+d\,x]^j)^m\,(A+B\,\text{Sin}[c+d\,x]^k)\,(a+b\,\text{Sin}[c+d\,x]^k)^n\,dx\,\longrightarrow\,\frac{B}{b}\int (\text{Sin}[c+d\,x]^j)^m\,(a+b\,\text{Sin}[c+d\,x]^k)^{n+1}\,dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(A_.+B_.*sin[c_.+d_.*x_]^k_.)*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol
  Dist[B/b,Int[(sin[c+d*x]^j)^m*(a+b*sin[c+d*x]^k)^(n+1),x]] /;
FreeQ[{a,b,c,d,A,B,m},x] && OneQ[j^2,k^2] && ZeroQ[b*A-a*B] && RationalQ[n] && n<0
```

- **Derivation: Recurrence 1 with $A = 0$, $B = A$, $C = B$ and $m = m - 1$**

- **Rule 1a: If $j^2 = k^2 = 1 \wedge a^2 - b^2 \neq 0 \wedge b A - a B \neq 0 \wedge j k m > 1 \wedge n < -1$, then**

$$\int \left(\operatorname{Sin}[c + d x]^j\right)^m \left(A + B \operatorname{Sin}[c + d x]^k\right) \left(a + b \operatorname{Sin}[c + d x]^k\right)^n dx \longrightarrow$$

$$\frac{a (b A - a B) \operatorname{Cos}[c + d x] \left(\operatorname{Sin}[c + d x]^j\right)^{m-j k} \left(a + b \operatorname{Sin}[c + d x]^k\right)^{n+1}}{b d (n + 1) \left(a^2 - b^2\right)} -$$

$$\frac{1}{b (n + 1) \left(a^2 - b^2\right)} \int \left(\operatorname{Sin}[c + d x]^j\right)^{m-2 j k} \cdot$$

$$\left(a (b A - a B) \left(j k m + \frac{k - 3}{2}\right) + b (b A - a B) (n + 1) \operatorname{Sin}[c + d x]^k -\right.$$

$$\left.\left(b (a A - b B) (n + 1) + a (b A - a B) \left(j k m + \frac{k - 1}{2}\right)\right) \operatorname{Sin}[c + d x]^{2 k}\right) \cdot$$

$$\left(a + b \operatorname{Sin}[c + d x]^k\right)^{n+1}$$

$$dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_*(A_.+B_.*sin[c_.+d_.*x_]^k_.)*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  a*(b*A-a*B)*Cos[c+d*x]*(Sin[c+d*x]^j)^(m-j*k)*(a+b*Sin[c+d*x]^k)^(n+1)/(b*d*(n+1)*(a^2-b^2)) -
  Dist[1/(b*(n+1)*(a^2-b^2)),
    Int[(sin[c+d*x]^j)^(m-2*j*k)*
      Sim[a*(b*A-a*B)*(j*k*m+(k-3)/2)+b*(b*A-a*B)*(n+1)*sin[c+d*x]^k-
        (b*(a*A-b*B)*(n+1)+a*(b*A-a*B)*(j*k*m+(k-1)/2))*sin[c+d*x]^(2*k),x]*
      (a+b*sin[c+d*x]^k)^(n+1),x]] /;
FreeQ[{a,b,c,d,A,B},x] && OneQ[j^2,k^2] && NonzeroQ[a^2-b^2] && NonzeroQ[b*A-a*B] &&
  RationalQ[m,n] && j*k*m>1 && n<-1
```

- **Derivation: Recurrence 1 with** $C = 0$
- **Derivation: Recurrence 6 with** $A = 0$, $B = A$, $C = B$ **and** $m = m - 1$
- **Rule 1b: If** $j^2 = k^2 = 1 \wedge a^2 - b^2 \neq 0 \wedge b A - a B \neq 0 \wedge 0 < j k m < 1 \wedge n < -1$, **then**

$$\int \left(\text{Sin}[c + d x]^j\right)^m \left(A + B \,\text{Sin}[c + d x]^k\right) \left(a + b \,\text{Sin}[c + d x]^k\right)^n dx \longrightarrow$$

$$-\frac{(b A - a B) \,\text{Cos}[c + d x] \left(\text{Sin}[c + d x]^j\right)^m \left(a + b \,\text{Sin}[c + d x]^k\right)^{n+1}}{d (n + 1) \left(a^2 - b^2\right)} +$$

$$\frac{1}{(n + 1) \left(a^2 - b^2\right)} \int \left(\text{Sin}[c + d x]^j\right)^{m-j k} \cdot$$

$$\left((b A - a B) \left(j k m + \frac{k - 1}{2}\right) + (a A - b B) (n + 1) \,\text{Sin}[c + d x]^k - (b A - a B) \left(j k m + n + \frac{k + 3}{2}\right) \text{Sin}[c + d x]^{2 k}\right) \cdot$$

$$\left(a + b \,\text{Sin}[c + d x]^k\right)^{n+1} dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_*(A_.+B_.*sin[c_.+d_.*x_]^k_.)*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  -(b*A-a*B)*Cos[c+d*x]*(Sin[c+d*x]^j)^m*(a+b*Sin[c+d*x]^k)^(n+1)/(d*(n+1)*(a^2-b^2)) +
  Dist[1/((n+1)*(a^2-b^2)),
    Int[(sin[c+d*x]^j)^(m-j*k)*
      Sim[(b*A-a*B)*(j*k*m+(k-1)/2)+(a*A-b*B)*(n+1)*sin[c+d*x]^k-
        (b*A-a*B)*(j*k*m+n+(k+3)/2)*sin[c+d*x]^(2*k),x]*
      (a+b*sin[c+d*x]^k)^(n+1),x]] /;
FreeQ[{a,b,c,d,A,B},x] && OneQ[j^2,k^2] && NonzeroQ[a^2-b^2] && NonzeroQ[b*A-a*B] &&
  RationalQ[m,n] && 0<j*k*m<1 && n<-1
```

- **Derivation: Recurrence 2 with $A = 0$, $B = A$, $C = B$ and $m = m - 1$**

- **Rule 2: If $j^2 = k^2 = 1 \wedge a^2 - b^2 \neq 0 \wedge j\,k\,m > 1 \wedge -1 \leq n < 0$, then**

$$\int \left(\text{Sin}[c+d\,x]^j\right)^m \left(A + B\,\text{Sin}[c+d\,x]^k\right) \left(a + b\,\text{Sin}[c+d\,x]^k\right)^n dx \longrightarrow$$

$$-\frac{B\,\text{Cos}[c+d\,x] \left(\text{Sin}[c+d\,x]^j\right)^{m-j\,k} \left(a + b\,\text{Sin}[c+d\,x]^k\right)^{n+1}}{b\,d \left(j\,k\,m + n + \frac{k+1}{2}\right)} +$$

$$\frac{1}{b \left(j\,k\,m + n + \frac{k+1}{2}\right)} \int \left(\text{Sin}[c+d\,x]^j\right)^{m-2\,j\,k} \cdot$$

$$\left(a\,B \left(j\,k\,m + \frac{k-3}{2}\right) + b\,B \left(j\,k\,m + n + \frac{k-1}{2}\right) \text{Sin}[c+d\,x]^k + \right.$$

$$\left. \left(b\,A\,(n+1) + (b\,A - a\,B) \left(j\,k\,m + \frac{k-1}{2}\right)\right) \text{Sin}[c+d\,x]^{2\,k}\right) \cdot$$

$$\left(a + b\,\text{Sin}[c+d\,x]^k\right)^n$$

$$dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_*(A_.+B_.*sin[c_.+d_.*x_]^k_.)*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  -B*Cos[c+d*x]*(Sin[c+d*x]^j)^(m-j*k)*(a+b*Sin[c+d*x]^k)^(n+1)/(b*d*(j*k*m+n+(k+1)/2)) +
  Dist[1/(b*(j*k*m+n+(k+1)/2)),
    Int[(sin[c+d*x]^j)^(m-2*j*k)*
      Sim[a*B*(j*k*m+(k-3)/2)+b*B*(j*k*m+n+(k-1)/2)*sin[c+d*x]^k+
        (b*A*(n+1)+(b*A-a*B)*(j*k*m+(k-1)/2))*sin[c+d*x]^(2*k),x]*
      (a+b*sin[c+d*x]^k)^n,x]] /;
FreeQ[{a,b,c,d,A,B},x] && OneQ[j^2,k^2] && NonzeroQ[a^2-b^2] && RationalQ[m,n] &&
  j*k*m>1 && -1≤n<0
```

- **Derivation:** Recurrence 3 with $A = a\,A$, $B = b\,A + a\,B$, $C = b\,B$ and $n = n - 1$
- **Rule 3a:** If $j^2 = k^2 = 1 \land a^2 - b^2 \neq 0 \land j\,k\,m \geq -1 \land j\,k\,m \neq 1 \land n > 1$, then

$$\int \left(\text{Sin}[c+d\,x]^j\right)^m \left(A + B\,\text{Sin}[c+d\,x]^k\right) \left(a + b\,\text{Sin}[c+d\,x]^k\right)^n dx \longrightarrow$$

$$-\frac{b\,B\,\text{Cos}[c+d\,x] \left(\text{Sin}[c+d\,x]^j\right)^{m+j\,k} \left(a + b\,\text{Sin}[c+d\,x]^k\right)^{n-1}}{d\left(j\,k\,m + n + \frac{k+1}{2}\right)} +$$

$$\frac{1}{j\,k\,m + n + \frac{k+1}{2}} \int \left(\text{Sin}[c+d\,x]^j\right)^m \cdot$$

$$\left(a\left(a\,A\,n + (a\,A + b\,B)\left(j\,k\,m + \frac{k+1}{2}\right)\right) + \left(a\,(2\,b\,A + a\,B) + \left(a^2\,B + 2\,a\,b\,A + b^2\,B\right)\left(j\,k\,m + n + \frac{k-1}{2}\right)\right)\right.$$

$$\left.\text{Sin}[c+d\,x]^k + b\left(a\,B\,(n-1) + (b\,A + a\,B)\left(j\,k\,m + n + \frac{k+1}{2}\right)\right)\text{Sin}[c+d\,x]^{2\,k}\right) \cdot$$

$$\left(a + b\,\text{Sin}[c+d\,x]^k\right)^{n-2}$$

$$dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(A_.+B_.*sin[c_.+d_.*x_]^k_.)*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_.,x_Symbo
  -b*B*Cos[c+d*x]*(Sin[c+d*x]^j)^(m+j*k)*(a+b*Sin[c+d*x]^k)^(n-1)/(d*(j*k*m+n+(k+1)/2)) +
  Dist[1/(j*k*m+n+(k+1)/2),
    Int[(sin[c+d*x]^j)^m*
      Sim[a*(a*A*n+(a*A+b*B)*(j*k*m+(k+1)/2))+
        (a*(2*b*A+a*B)+(a^2*B+2*a*b*A+b^2*B)*(j*k*m+n+(k-1)/2))*sin[c+d*x]^k+
         b*(a*B*(n-1)+(b*A+a*B)*(j*k*m+n+(k+1)/2))*sin[c+d*x]^(2*k),x]*
      (a+b*sin[c+d*x]^k)^(n-2),x]] /;
FreeQ[{a,b,c,d,A,B},x] && OneQ[j^2,k^2] && NonzeroQ[a^2-b^2] && RationalQ[m,n] &&
  j*k*m≥-1 && j*k*m≠1 && n>1
```

- **Derivation: Recurrence 2 with** $A = a A$, $B = b A + a B$, $C = b B$ **and** $n = n - 1$
- **Derivation: Recurrence 3 with** $A = 0$, $B = A$, $C = B$ **and** $m = m - 1$
- **Rule 3b: If** $j^2 = k^2 = 1 \land a^2 - b^2 \neq 0 \land j k m > 0 \land j k m \neq 1 \land 0 < n < 1$, **then**

$$\int \left(\operatorname{Sin}[c+d x]^j\right)^m \left(A+B \operatorname{Sin}[c+d x]^k\right) \left(a+b \operatorname{Sin}[c+d x]^k\right)^n dx \longrightarrow$$

$$-\frac{B \operatorname{Cos}[c+d x] \left(\operatorname{Sin}[c+d x]^j\right)^m \left(a+b \operatorname{Sin}[c+d x]^k\right)^n}{d \left(j k m+n+\frac{k+1}{2}\right)}+$$

$$\frac{1}{j k m+n+\frac{k+1}{2}} \int \left(\operatorname{Sin}[c+d x]^j\right)^{m-j k} \cdot$$

$$\left(a B \left(j k m+\frac{k-1}{2}\right)+\left(a A+(a A+b B)\left(j k m+n+\frac{k-1}{2}\right)\right) \operatorname{Sin}[c+d x]^k+\right.$$

$$\left.\left(n (b A+a B)+b A \left(j k m+\frac{k+1}{2}\right)\right) \operatorname{Sin}[c+d x]^{2 k}\right) \cdot$$

$$\left(a+b \operatorname{Sin}[c+d x]^k\right)^{n-1}$$

$$dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(A_.+B_.*sin[c_.+d_.*x_]^k_.)*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_.,x_Symbo
  -B*Cos[c+d*x]*(Sin[c+d*x]^j)^m*(a+b*Sin[c+d*x]^k)^n/(d*(j*k*m+n+(k+1)/2)) +
  Dist[1/(j*k*m+n+(k+1)/2),
    Int[(sin[c+d*x]^j)^(m-j*k)*
      Sim[a*B*(j*k*m+(k-1)/2)+(a*A+(a*A+b*B)*(j*k*m+n+(k-1)/2))*sin[c+d*x]^k+
        (n*(b*A+a*B)+b*A*(j*k*m+(k+1)/2))*sin[c+d*x]^(2*k),x]*
      (a+b*sin[c+d*x]^k)^(n-1),x]] /;
FreeQ[{a,b,c,d,A,B},x] && OneQ[j^2,k^2] && NonzeroQ[a^2-b^2] && RationalQ[m,n] &&
  j*k*m>0 && j*k*m≠1 && 0<n<1
```

- **Derivation:** Recurrence 4 with $A = a A$, $B = b A + a B$, $C = b B$ and $n = n - 1$
- **Rule 4a:** If $j^2 = k^2 = 1 \wedge a^2 - b^2 \neq 0 \wedge j k m < -1 \wedge n > 1$, then

$$\int \left(\text{Sin}[c+d\,x]^j\right)^m \left(A + B\,\text{Sin}[c+d\,x]^k\right)\left(a+b\,\text{Sin}[c+d\,x]^k\right)^n dx \longrightarrow$$

$$\frac{a\,A\,\text{Cos}[c+d\,x]\left(\text{Sin}[c+d\,x]^j\right)^{m+j\,k}\left(a+b\,\text{Sin}[c+d\,x]^k\right)^{n-1}}{d\left(j\,k\,m+\frac{k+1}{2}\right)} +$$

$$\frac{1}{j\,k\,m+\frac{k+1}{2}}\int\left(\text{Sin}[c+d\,x]^j\right)^{m+j\,k}\cdot$$

$$\left(a\left((b\,A+a\,B)\left(j\,k\,m+\frac{k+1}{2}\right)-b\,A\,(n-1)\right)+\left(a^2\,A+\left(a^2\,A+2\,a\,b\,B+b^2\,A\right)\left(j\,k\,m+\frac{k+1}{2}\right)\right)\text{Sin}[c+d\,x]^k+\right.$$

$$\left.b\left(a\,A\,n+(a\,A+b\,B)\left(j\,k\,m+\frac{k+1}{2}\right)\right)\text{Sin}[c+d\,x]^{2\,k}\right)\cdot$$

$$\left(a+b\,\text{Sin}[c+d\,x]^k\right)^{n-2}$$

$$dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(A_.+B_.*sin[c_.+d_.*x_]^k_.)*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol
  a*A*Cos[c+d*x]*(Sin[c+d*x]^j)^(m+j*k)*(a+b*Sin[c+d*x]^k)^(n-1)/(d*(j*k*m+(k+1)/2)) +
  Dist[1/(j*k*m+(k+1)/2),
    Int[(sin[c+d*x]^j)^(m+j*k)*
      Sim[a*((b*A+a*B)*(j*k*m+(k+1)/2)-b*A*(n-1))+
        (a^2*A+(a^2*A+2*a*b*B+b^2*A)*(j*k*m+(k+1)/2))*sin[c+d*x]^k+
        b*(a*A*n+(a*A+b*B)*(j*k*m+(k+1)/2))*sin[c+d*x]^(2*k),x]*
      (a+b*sin[c+d*x]^k)^(n-2),x]] /;
FreeQ[{a,b,c,d,A,B},x] && OneQ[j^2,k^2] && NonzeroQ[a^2-b^2] && RationalQ[m,n] &&
  j*k*m<-1 && n>1
```

- **Derivation: Recurrence 4 with** $C = 0$

- **Rule 4b: If** $j^2 = k^2 = 1 \land a^2 - b^2 \neq 0 \land j\,k\,m < -1 \land 0 < n \leq 1$, **then**

$$\int \left(\text{Sin}[c+d\,x]^j\right)^m \left(A + B\,\text{Sin}[c+d\,x]^k\right) \left(a + b\,\text{Sin}[c+d\,x]^k\right)^n dx \longrightarrow$$

$$\frac{A\,\text{Cos}[c+d\,x] \left(\text{Sin}[c+d\,x]^j\right)^{m+j\,k} \left(a + b\,\text{Sin}[c+d\,x]^k\right)^n}{d\left(j\,k\,m + \frac{k+1}{2}\right)} +$$

$$\frac{1}{j\,k\,m + \frac{k+1}{2}} \int \left(\text{Sin}[c+d\,x]^j\right)^{m+j\,k} \cdot$$

$$\left(a\,B\left(j\,k\,m + \frac{k+1}{2}\right) - b\,A\,n +\right.$$

$$\left.\left(a\,A + (a\,A + b\,B)\left(j\,k\,m + \frac{k+1}{2}\right)\right) \text{Sin}[c+d\,x]^k + b\,A\left(j\,k\,m + n + \frac{k+3}{2}\right) \text{Sin}[c+d\,x]^{2\,k}\right) \cdot$$

$$\left(a + b\,\text{Sin}[c+d\,x]^k\right)^{n-1}$$

$$dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(A_.+B_.*sin[c_.+d_.*x_]^k_.)*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_.,x_Symbo
  A*Cos[c+d*x]*(Sin[c+d*x]^j)^(m+j*k)*(a+b*Sin[c+d*x]^k)^n/(d*(j*k*m+(k+1)/2)) +
  Dist[1/(j*k*m+(k+1)/2),
    Int[(sin[c+d*x]^j)^(m+j*k)*
      Sim[a*B*(j*k*m+(k+1)/2)-b*A*n+(a*A+(a*A+b*B)*(j*k*m+(k+1)/2))*sin[c+d*x]^k+
        b*A*(j*k*m+n+(k+3)/2)*sin[c+d*x]^(2*k),x]*
      (a+b*sin[c+d*x]^k)^(n-1),x]] /;
FreeQ[{a,b,c,d,A,B},x] && OneQ[j^2,k^2] && NonzeroQ[a^2-b^2] && RationalQ[m,n] &&
  j*k*m<-1 && 0<n≤1
```

- **Derivation: Recurrence 5 with** $C = 0$

- **Rule 5: If** $j^2 = k^2 = 1 \bigwedge a^2 - b^2 \neq 0 \bigwedge j\,k\,m + \frac{k+1}{2} \neq 0 \bigwedge j\,k\,m \leq -1 \bigwedge -1 \leq n < 0$**, then**

$$\int \left(\mathrm{Sin}[c + d\,x]^j\right)^m \left(A + B\,\mathrm{Sin}[c + d\,x]^k\right) \left(a + b\,\mathrm{Sin}[c + d\,x]^k\right)^n \,dx \longrightarrow$$

$$\frac{A\,\mathrm{Cos}[c + d\,x] \left(\mathrm{Sin}[c + d\,x]^j\right)^{m+j\,k} \left(a + b\,\mathrm{Sin}[c + d\,x]^k\right)^{n+1}}{a\,d \left(j\,k\,m + \frac{k+1}{2}\right)} +$$

$$\frac{1}{a \left(j\,k\,m + \frac{k+1}{2}\right)} \int \left(\mathrm{Sin}[c + d\,x]^j\right)^{m+j\,k} \cdot$$

$$\left((a\,B - b\,A) \left(j\,k\,m + \frac{k+1}{2}\right) - b\,A\,(n + 1) + \right.$$

$$\left. a\,A \left(j\,k\,m + \frac{k+3}{2}\right) \mathrm{Sin}[c + d\,x]^k + b\,A \left(j\,k\,m + n + \frac{k+3}{2}\right) \mathrm{Sin}[c + d\,x]^{2\,k} \right) \cdot$$

$$\left(a + b\,\mathrm{Sin}[c + d\,x]^k\right)^n$$

$$dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(A_.+B_.*sin[c_.+d_.*x_]^k_.)*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol
  A*Cos[c+d*x]*(Sin[c+d*x]^j)^(m+j*k)*(a+b*Sin[c+d*x]^k)^(n+1)/(a*d*(j*k*m+(k+1)/2)) +
  Dist[1/(a*(j*k*m+(k+1)/2)),
    Int[(sin[c+d*x]^j)^(m+j*k)*
      Sim[(a*B-b*A)*(j*k*m+(k+1)/2)-b*A*(n+1)+a*A*(j*k*m+(k+3)/2)*sin[c+d*x]^k+
        b*A*(j*k*m+n+(k+1)/2+2)*sin[c+d*x]^(2*k),x]*
      (a+b*sin[c+d*x]^k)^n,x]] /;
FreeQ[{a,b,c,d,A,B},x] && OneQ[j^2,k^2] && NonzeroQ[a^2-b^2] && RationalQ[m,n] &&
  j*k*m+(k+1)/2≠0 && j*k*m≤-1 && -1≤n<0
```

- **Derivation: Recurrence 6 with** $C = 0$
- **Rule 6: If** $j^2 = k^2 = 1 \land a^2 - b^2 \neq 0 \land b A - a B \neq 0 \land j k m < 0 \land n < -1$**, then**

$$\int \left(\text{Sin}[c + d x]^j\right)^m \left(A + B\,\text{Sin}[c + d x]^k\right) \left(a + b\,\text{Sin}[c + d x]^k\right)^n dx \longrightarrow$$

$$\frac{b\,(b A - a B)\,\text{Cos}[c + d x] \left(\text{Sin}[c + d x]^j\right)^{m+j k} \left(a + b\,\text{Sin}[c + d x]^k\right)^{n+1}}{a d\,(n + 1) \left(a^2 - b^2\right)} +$$

$$\frac{1}{a\,(n + 1) \left(a^2 - b^2\right)} \int \left(\text{Sin}[c + d x]^j\right)^m \cdot$$

$$\left(A \left(a^2 - b^2\right) (n + 1) - b\,(b A - a B) \left(j k m + \frac{k + 1}{2}\right) - \right.$$

$$\left. a\,(b A - a B)\,(n + 1)\,\text{Sin}[c + d x]^k + b\,(b A - a B) \left(j k m + n + \frac{k + 5}{2}\right) \text{Sin}[c + d x]^{2 k}\right) \cdot$$

$$\left(a + b\,\text{Sin}[c + d x]^k\right)^{n+1}$$

$$dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(A_.+B_.*sin[c_.+d_.*x_]^k_.)*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol
  b*(b*A-a*B)*Cos[c+d*x]*(Sin[c+d*x]^j)^(m+j*k)*(a+b*Sin[c+d*x]^k)^(n+1)/(a*d*(n+1)*(a^2-b^2)) +
  Dist[1/(a*(n+1)*(a^2-b^2)),
    Int[(sin[c+d*x]^j)^m*
      Sim[A*(a^2-b^2)*(n+1)-b*(b*A-a*B)*(j*k*m+(k+1)/2)-a*(b*A-a*B)*(n+1)*sin[c+d*x]^k+
        b*(b*A-a*B)*(j*k*m+n+(k+5)/2)*sin[c+d*x]^(2*k),x]*
      (a+b*sin[c+d*x]^k)^(n+1),x]] /;
FreeQ[{a,b,c,d,A,B},x] && OneQ[j^2,k^2] && NonzeroQ[a^2-b^2] && NonzeroQ[b*A-a*B] &&
  RationalQ[m,n] && j*k*m<0 && n<-1
```