$\int a\,u\,dx$

- **Reference: CRC 1**
- **Rule:**

$$\int a\,dx \longrightarrow a\,x$$

- **Program code:**

```
Int[a_,x_Symbol] :=
   a*x /;
IndependentQ[a,x]
```

- **Derivation: Power rule for integration**
- **Rule:**

$$\int c\,(a+b\,x)\,dx \longrightarrow \frac{c\,(a+b\,x)^2}{2\,b}$$

- **Program code:**

```
Int[c_*(a_+b_.*x_),x_Symbol] :=
  c*(a+b*x)^2/(2*b) /;
FreeQ[{a,b,c},x]
```

- **Reference: G&R 2.02.1, CRC 2**
- **Derivation: Constant extraction**
- **Rule: If $n+1\neq 0$, then**

$$\int c\,(a+b\,x)^n\,dx \longrightarrow c\int (a+b\,x)^n\,dx$$

- **Program code:**

```
Int[c_*(a_+b_.*x_)^n_,x_Symbol] :=
  Dist[c,Int[(a+b*x)^n,x]] /;
FreeQ[{a,b,c,n},x] && NonzeroQ[n+1]
```

- **Reference: G&R 2.02.1, CRC 2**
- **Derivation: Constant extraction**
- **Rule:**

$$\int a\,u\,dx \longrightarrow a\int u\,dx$$

- **Program code:**

```
If[ShowSteps,

Int[a_*u_,x_Symbol] :=
  Module[{lst=ConstantFactor[u,x]},
  ShowStep["","Int[a*u,x]","a*Int[u,x]",Hold[
  Dist[a*lst[[1]],Int[lst[[2]],x]]]]] /;
SimplifyFlag && FreeQ[a,x] && Not[MatchQ[u,b_*v_ /; FreeQ[b,x]]],

Int[a_*u_,x_Symbol] :=
  Module[{lst=ConstantFactor[u,x]},
  Dist[a*lst[[1]],Int[lst[[2]],x]]] /;
FreeQ[a,x] && Not[MatchQ[u,b_*v_ /; FreeQ[b,x]]]]
```

- **Derivation: Constant extraction**
- **Note: Constant factors in denominators are aggressively factored out to prevent them occurring unnecessarily in logarithm terms of antiderivatives!**
- **Rule:**

$$\int a\,u\,dx \longrightarrow a\int u\,dx$$

- **Program code:**

```
If[ShowSteps,

Int[u_,x_Symbol] :=
  Module[{lst=ConstantFactor[Simplify[Denominator[u]],x]},
  ShowStep["","Int[a*u,x]","a*Int[u,x]",Hold[
  Dist[1/lst[[1]],Int[Numerator[u]/lst[[2]],x]]]] /;
 lst[[1]]=!=1] /;
SimplifyFlag && (
    MatchQ[u,1/(a_+b_.*x) /; FreeQ[{a,b},x]] ||
    MatchQ[u,x^m_./(a_+b_.*x^n_) /; FreeQ[{a,b,m,n},x] && ZeroQ[m-n+1]] ||
    MatchQ[u,1/((a_.+b_.*x)*(c_+d_.*x)) /; FreeQ[{a,b,c,d},x]] ||
    MatchQ[u,(d_.+e_.*x)/(a_+b_.*x+c_.*x^2) /; FreeQ[{a,b,c,d,e},x]] ||
    MatchQ[u,(c_.*(a_.+b_.*x)^n_)^m_ /; FreeQ[{a,b,c,m,n},x] && ZeroQ[m*n+1]]),

Int[u_,x_Symbol] :=
  Module[{lst=ConstantFactor[Simplify[Denominator[u]],x]},
  Dist[1/lst[[1]],Int[Numerator[u]/lst[[2]],x]] /;
 lst[[1]]=!=1] /;
    MatchQ[u,1/(a_+b_.*x) /; FreeQ[{a,b},x]] ||
    MatchQ[u,x^m_./(a_+b_.*x^n_) /; FreeQ[{a,b,m,n},x] && ZeroQ[m-n+1]] ||
    MatchQ[u,1/((a_.+b_.*x)*(c_+d_.*x)) /; FreeQ[{a,b,c,d},x]] ||
    MatchQ[u,(d_.+e_.*x)/(a_+b_.*x+c_.*x^2) /; FreeQ[{a,b,c,d,e},x]] ||
    MatchQ[u,(c_.*(a_.+b_.*x)^n_)^m_ /; FreeQ[{a,b,c,m,n},x] && ZeroQ[m*n+1]]]
```

- **Derivation: Constant extraction**

- **Note: Constant factors in denominators are aggressively factored out to prevent them occurring unnecessarily in logarithm terms of antiderivatives!**

- **Rule:**

$$\int a\,u\,dx \longrightarrow a\int u\,dx$$

- **Program code:**

```
If[ShowSteps,

Int[u_/v_,x_Symbol] :=
  Module[{lst=ConstantFactor[v,x]},
  ShowStep["","Int[a*u,x]","a*Int[u,x]",Hold[
  Dist[1/lst[[1]],Int[u/lst[[2]],x]]]]] /;
 lst[[1]]=!=1] /;
SimplifyFlag && Not[FalseQ[DerivativeDivides[v,u,x]]],

Int[u_/v_,x_Symbol] :=
  Module[{lst=ConstantFactor[v,x]},
  Dist[1/lst[[1]],Int[u/lst[[2]],x]] /;
 lst[[1]]=!=1] /;
Not[FalseQ[DerivativeDivides[v,u,x]]]]
```

- **Derivation: Piecewise constant extraction**

- **Basis:** $\partial_x \frac{f[x]^m}{(-f[x])^m} = 0$

- **Rule: If** $m + n = 0 \wedge v + w = 0$

$$\int u\,v^m\,w^n\,dx \longrightarrow v^m\,w^n\int u\,dx$$

- **Program code:**

```
Int[u_.*v_^m_*w_^n_,x_Symbol] :=
  (v^m*w^n)*Int[u,x] /;
FreeQ[{m,n},x] && ZeroQ[m+n] && ZeroQ[v+w]
```

- **Derivation: Piecewise constant extraction**
- **Basis:** $\partial_x \frac{(a+b\,x^m)^p}{x^{m\,p}\left(-b-\frac{a}{x^m}\right)^p} = 0$
- **Rule: If** $a + d = 0\ \wedge\ b + c = 0\ \wedge\ m + n = 0\ \wedge\ p + q = 0$

$$\int u\,(a + b\,x^m)^p\,(c + d\,x^n)^q\,dx \longrightarrow \frac{(a + b\,x^m)^p\,(c + d\,x^n)^q}{x^{m\,p}} \int u\,x^{m\,p}\,dx$$

- **Program code:**

```
Int[u_.*(a_.+b_.*x_^m_.)^p_.*(c_.+d_.*x_^n_.)^q_., x_Symbol] :=
  (a+b*x^m)^p*(c+d*x^n)^q/x^(m*p)*Int[u*x^(m*p),x] /;
FreeQ[{a,b,c,d,m,n,p,q},x] && ZeroQ[a+d] && ZeroQ[b+c] && ZeroQ[m+n] && ZeroQ[p+q]
```

$$\int (a + b\,x)^n\,dx$$

- **Reference: G&R 2.01.2, CRC 27, A&S 3.3.15**
- **Derivation: Reciprocal rule for integration**
- **Rule:**

$$\int \frac{1}{-a + b\,x}\,dx \longrightarrow \frac{\text{Log}[a - b\,x]}{b}$$

- **Program code:**

```
Int[1/(a_+b_.*x_),x_Symbol] :=
  Log[-a-b*x]/b /;
FreeQ[{a,b},x] && NegativeCoefficientQ[a]
```

- **Reference: G&R 2.01.2, CRC 27, A&S 3.3.15**
- **Derivation: Reciprocal rule for integration**
- **Rule:**

$$\int \frac{1}{a + b\,x}\,dx \longrightarrow \frac{\text{Log}[a + b\,x]}{b}$$

- **Program code:**

```
Int[1/(a_.+b_.*x_),x_Symbol] :=
  Log[a+b*x]/b /;
FreeQ[{a,b},x]
```

- **Reference: G&R 2.01.1, CRC 7**
- **Derivation: Power rule for integration**
- **Rule: If $n + 1 \neq 0$, then**

$$\int x^n\,dx \longrightarrow \frac{x^{n+1}}{n+1}$$

- **Program code:**

```
Int[x_^n_.,x_Symbol] :=
  x^(n+1)/(n+1) /;
IndependentQ[n,x] && NonzeroQ[n+1]
```

- **Reference: G&R 2.01.1, CRC 23, A&S 3.3.14**
- **Derivation: Power rule for integration**
- **Rule: If $n+1 \neq 0$, then**

$$\int (a+b\,x)^n\,dx \longrightarrow \frac{(a+b\,x)^{n+1}}{b\,(n+1)}$$

- **Program code:**

```
Int[(a_.+b_.*x_)^n_,x_Symbol] :=
  (a+b*x)^(n+1)/(b*(n+1)) /;
FreeQ[{a,b,n},x] && NonzeroQ[n+1]
```

$$\int a\,x^m + b\,x^n + \cdots\,dx$$

- **Reference: CRC 1,2,4,7,9**
- **Rule:**

$$\int a + b\,x + c\,x^2 + \cdots\,dx \;\longrightarrow\; a\,x + \frac{b\,x^2}{2} + \frac{c\,x^3}{3} + \cdots$$

- **Program code:**

```
If[ShowSteps,

Int[u_,x_Symbol] :=
  If[PolynomialQ[u,x],
    ShowStep["","Int[a+b*x+c*x^2+⋯,x]","a*x+b*x^2/2+c*x^3/3+⋯",Hold[
    IntegrateMonomialSum[u,x]]],
  ShowStep["","Int[a+b/x+c*x^m+⋯,x]","a*x+b*Log[x]+c*x^(m+1)/(m+1)+⋯",Hold[
  IntegrateMonomialSum[u,x]]]] /;
SimplifyFlag && MonomialSumQ[u,x],

Int[u_,x_Symbol] :=
  IntegrateMonomialSum[u,x] /;
MonomialSumQ[u,x]]
```

- **Reference: G&R 2.02.2, CRC 2,4**
- **Rule:**

$$\int a\,u + b\,v + \cdots\,dx \;\longrightarrow\; a\int u\,dx + b\int v\,dx + \cdots$$

- **Program code:**

```
If[ShowSteps,

Int[u_,x_Symbol] :=
  Module[{lst=SplitMonomialTerms[u,x]},
  ShowStep["","Int[a*u+b*v+⋯,x]","a*Int[u,x]+b*Int[v,x]+⋯",Hold[
  Int[lst[[1]],x] + SplitFreeIntegrate[lst[[2]],x]]] /;
 SumQ[lst[[1]]] && Not[FreeQ[lst[[1]],x]] && lst[[2]]=!=0] /;
SimplifyFlag && SumQ[u],

Int[u_,x_Symbol] :=
  Module[{lst=SplitMonomialTerms[u,x]},
  Int[lst[[1]],x] + SplitFreeIntegrate[lst[[2]],x] /;
 SumQ[lst[[1]]] && Not[FreeQ[lst[[1]],x]] && lst[[2]]=!=0] /;
SumQ[u]]
```

- **Derivation: Algebraic expansion**
- **Basis:** $z\,(u+v+\cdots) = z\,u + z\,v + \cdots$
- **Rule: If** $m \in \mathbb{Z}$, **then**

$$\int x^m\,(u+v+\cdots)\,dx \;\longrightarrow\; \int x^m\,u + x^m\,v + \cdots\,dx$$

- **Program code:**

```
If[ShowSteps,

Int[x_^m_.*u_,x_Symbol] :=
  ShowStep["","Int[x^m*(u+v+⋯),x]","Int[x^m*u+x^m*v+⋯,x]",Hold[
  Int[Map[Function[x^m*#],u],x]]] /;
SimplifyFlag && IntegerQ[m] && SumQ[u],

Int[x_^m_.*u_,x_Symbol] :=
  Int[Map[Function[x^m*#],u],x] /;
IntegerQ[m] && SumQ[u]]
```

$$\int \frac{1}{x\,(a+b\,x^n)}\,dx$$

- **Derivation: Integration by substitution**
- **Basis: If** $u = 1 + \frac{2\,b\,x^n}{a}$**, then** $\frac{1}{x\,(a+b\,x^n)} = -\frac{2}{a\,n}\,\frac{1}{1-u^2}\,\partial_x u$
- **Rule: If** $n > 0 \bigwedge a \in \mathbb{Q}$**, then**

$$\int \frac{1}{x\,(a+b\,x^n)}\,dx \longrightarrow -\frac{2}{a\,n}\,\text{ArcTanh}\left[1+\frac{2\,b\,x^n}{a}\right]$$

- **Program code:**

```
Int[1/(x_*(a_+b_.*x_^n_.)),x_Symbol] :=
  -2*ArcTanh[1+2*b*x^n/a]/(a*n) /;
FreeQ[{a,b,n},x] && PosQ[n] && (RationalQ[a] || RationalQ[b/a])
```

- **Reference: G&R 2.118.1, CRC 84**
- **Derivation: Algebraic expansion and reciprocal rule for integration**
- **Basis:** $\frac{1}{x\,(a+b\,x^n)} = \frac{1}{a\,x} - \frac{b\,x^{n-1}}{a\,(a+b\,x^n)}$
- **Rule: If** $n > 0 \bigwedge \neg\,(a \in \mathbb{Q})$**, then**

$$\int \frac{1}{x\,(a+b\,x^n)}\,dx \longrightarrow \frac{\text{Log}[x]}{a} - \frac{\text{Log}[a+b\,x^n]}{a\,n}$$

- **Program code:**

```
Int[1/(x_*(a_+b_.*x_^n_.)),x_Symbol] :=
  Log[x]/a - Log[a+b*x^n]/(a*n) /;
FreeQ[{a,b,n},x] && PosQ[n] && Not[RationalQ[a] || RationalQ[b/a]]
```

- **Derivation: Reciprocal rule for integration**
- **Basis:** $\frac{1}{x\,(a+b\,x^n)} = \frac{1}{x^{n+1}\,\left(b+\frac{a}{x^n}\right)}$
- **Rule: If** $\neg\,(n>0)$**, then**

$$\int \frac{1}{x\,(a+b\,x^n)}\,dx \longrightarrow -\frac{\text{Log}[b+a\,x^{-n}]}{a\,n}$$

- **Program code:**

```
Int[1/(x_*(a_+b_.*x_^n_.)),x_Symbol] :=
  -Log[b+a*x^(-n)]/(a*n) /;
FreeQ[{a,b,n},x] && NegQ[n]
```

- **Derivation: Algebraic simplification**
- **Basis:** $a\,x+b\,x^n = x\,(a+b\,x^{n-1})$
- **Rule:**

$$\int \frac{1}{a\,x+b\,x^n}\,dx \longrightarrow \int \frac{1}{x\,\left(a+b\,x^{n-1}\right)}\,dx$$

- **Program code:**

```
Int[1/(a_.*x_+b_.*x_^n_),x_Symbol] :=
  Int[1/(x*(a+b*x^(n-1))),x] /;
FreeQ[{a,b,n},x]
```

$$\int x^m (a+b x)^n \, dx$$

- **Reference: G&R 2.110.2, CRC 26b special case with $m+n+2=0$**

- **Rule: If $m+n+2=0 \land n+1\neq 0$, then**

$$\int x^m (a+b x)^n \, dx \longrightarrow -\frac{x^{m+1} (a+b x)^{n+1}}{a (n+1)}$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*x_)^n_,x_Symbol] :=
  -x^(m+1)*(a+b*x)^(n+1)/(a*(n+1)) /;
FreeQ[{a,b,m,n},x] && ZeroQ[m+n+2] && NonzeroQ[n+1]
```

- **Reference: G&R 2.110.2, CRC 26b**

- **Derivation: Integration by parts**

- **Basis: $x^m (a+b x)^n = x^{m+n+2} \frac{(a+b x)^n}{x^{n+2}}$**

- **Rule: If $m, n \in \mathbb{Z} \land 0<m<-n-2 \land 2m+n+1>0$, then**

$$\int x^m (a+b x)^n \, dx \longrightarrow -\frac{x^{m+1} (a+b x)^{n+1}}{a (n+1)} + \frac{m+n+2}{a (n+1)} \int x^m (a+b x)^{n+1} \, dx$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*x_)^n_,x_Symbol] :=
  -x^(m+1)*(a+b*x)^(n+1)/(a*(n+1)) +
  Dist[(m+n+2)/(a*(n+1)),Int[x^m*(a+b*x)^(n+1),x]] /;
FreeQ[{a,b},x] && IntegersQ[m,n] && 0<m<-n-2 && 2*m+n+1>0
```

- **Reference: G&R 2.110.1, CRC 26a**
- **Derivation: Inverted integration by parts**
- **Rule: If $m, n \in \mathbb{Z} \wedge 0 < n < m/2$, then**

$$\int x^m (a+b x)^n \, dx \longrightarrow \frac{x^{m+1} (a+b x)^n}{m+n+1} + \frac{a n}{m+n+1} \int x^m (a+b x)^{n-1} \, dx$$

- **Program code:**

```
Int[x_^m_*(a_.+b_.*x_)^n_.,x_Symbol] :=
  x^(m+1)*(a+b*x)^n/(m+n+1) +
  Dist[a*n/(m+n+1),Int[x^m*(a+b*x)^(n-1),x]] /;
FreeQ[{a,b},x] && IntegersQ[m,n] && 0<n<m/2
```

- **Reference: G&R 2.110.6, CRC 88c**
- **Derivation: Integration by parts**
- **Basis: $x^m (a+b x)^n = \frac{x^m}{(a+b x)^{m+2}} (a+b x)^{m+n+2}$**
- **Rule: If $m, n \in \mathbb{Z} \wedge 0 < n < -m-2 \wedge m+2n-1 > 0$, then**

$$\int x^m (a+b x)^n \, dx \longrightarrow \frac{x^{m+1} (a+b x)^{n+1}}{a (m+1)} - \frac{b (m+n+2)}{a (m+1)} \int x^{m+1} (a+b x)^n \, dx$$

- **Program code:**

```
Int[x_^m_*(a_.+b_.*x_)^n_.,x_Symbol] :=
  x^(m+1)*(a+b*x)^(n+1)/(a*(m+1)) -
  Dist[b*(m+n+2)/(a*(m+1)),Int[x^(m+1)*(a+b*x)^n,x]] /;
FreeQ[{a,b},x] && IntegersQ[m,n] && 0<n<-m-2 && m+2*n-1>0
```

- **Reference: G&R 2.110.5, CRC 26c**
- **Derivation: Inverted integration by parts**
- **Rule: If $m, n \in \mathbb{Z} \wedge 0 < m < n/2$, then**

$$\int x^m (a+b x)^n \, dx \longrightarrow \frac{x^m (a+b x)^{n+1}}{b (m+n+1)} - \frac{a m}{b (m+n+1)} \int x^{m-1} (a+b x)^n \, dx$$

- **Program code:**

```
Int[x_^m_.*(a_.+b_.*x_)^n_,x_Symbol] :=
  x^m*(a+b*x)^(n+1)/(b*(m+n+1)) -
  Dist[a*m/(b*(m+n+1)),Int[x^(m-1)*(a+b*x)^n,x]] /;
FreeQ[{a,b},x] && IntegersQ[m,n] && 0<m<n/2
```

$$\int (a+b\,x)^m\,(c+d\,x)^n\,dx$$

- **Derivation: Algebraic simplification**
- **Basis: If** $b\,c+a\,d=0$**, then** $(a+b\,x)\,(c+d\,x)=a\,c+b\,d\,x^2$
- **Rule: If** $n\in\mathbb{Z}\,\wedge\,b\,c+a\,d=0$**, then**

$$\int (a+b\,x)^n\,(c+d\,x)^n\,dx\,\longrightarrow\,\int \left(a\,c+b\,d\,x^2\right)^n\,dx$$

- **Program code:**

```
Int[(a_+b_.*x_)^n_.*(c_+d_.*x_)^n_.,x_Symbol] :=
  Int[(a*c+b*d*x^2)^n,x] /;
FreeQ[{a,b,c,d},x] && IntegerQ[n] && ZeroQ[b*c+a*d]
```

- **Derivation: Algebraic simplification**
- **Basis: If** $b\,c-a\,d=0$ **and** $n\in\mathbb{Z}$**, then** $(a+b\,x)^m\,(c+d\,x)^n=\left(\frac{d}{b}\right)^n\,(a+b\,x)^{m+n}$
- **Rule: If** $b\,c-a\,d=0\,\wedge\,n\in\mathbb{Z}$**, then**

$$\int (a+b\,x)^m\,(c+d\,x)^n\,dx\,\longrightarrow\,\left(\frac{d}{b}\right)^n\int (a+b\,x)^{m+n}\,dx$$

- **Program code:**

```
Int[(a_+b_.*x_)^m_.*(c_+d_.*x_)^n_.,x_Symbol] :=
  Dist[(d/b)^n,Int[(a+b*x)^(m+n),x]] /;
FreeQ[{a,b,c,d,m},x] && ZeroQ[b*c-a*d] && IntegerQ[n] &&
(Not[IntegerQ[m]] || LeafCount[a+b*x]<=LeafCount[c+d*x])
```

- **Derivation: Integration by substitution**

- **Basis: If** $u = \frac{b\,c+a\,d}{b\,c-a\,d} + \frac{2\,b\,d\,x}{b\,c-a\,d}$**, then** $\frac{1}{(a+b\,x)\,(c+d\,x)} = -\frac{2}{b\,c-a\,d}\,\frac{1}{1-u^2}\,\partial_x u$

- **Note: If** $b\,c-a\,d \notin \mathbb{Q}$**, partial fraction expansion produces two nicer looking log terms.**

- **Rule: If** $b\,c-a\,d \in \mathbb{Q} \wedge b\,c-a\,d \neq 0$**, then**

$$\int \frac{1}{(a+b\,x)\,(c+d\,x)}\,dx \longrightarrow -\frac{2}{b\,c-a\,d}\,\text{ArcTanh}\left[\frac{b\,c+a\,d}{b\,c-a\,d}+\frac{2\,b\,d\,x}{b\,c-a\,d}\right]$$

- **Program code:**

```
Int[1/((a_+b_.*x_)*(c_+d_.*x_)),x_Symbol] :=
  -2*ArcTanh[(b*c+a*d)/(b*c-a*d) + 2*b*d*x/(b*c-a*d)]/(b*c-a*d) /;
FreeQ[{a,b,c,d},x] && RationalQ[b*c-a*d] && b*c-a*d!=0
```

- **Reference: G&R 2.155, CRC 59a special case when** $m+n+2=0$

- **Rule: If** $m+n+2=0 \wedge b\,c-a\,d \neq 0 \wedge n+1 \neq 0$**, then**

$$\int (a+b\,x)^m\,(c+d\,x)^n\,dx \longrightarrow -\frac{(a+b\,x)^{m+1}\,(c+d\,x)^{n+1}}{(n+1)\,(b\,c-a\,d)}$$

- **Program code:**

```
Int[(a_+b_.*x_)^m_.*(c_+d_.*x_)^n_,x_Symbol] :=
  -(a+b*x)^(m+1)*(c+d*x)^(n+1)/((n+1)*(b*c-a*d)) /;
FreeQ[{a,b,c,d,m,n},x] && ZeroQ[m+n+2] && NonzeroQ[b*c-a*d] && NonzeroQ[n+1]
```

- **Reference: G&R 2.155, CRC 59a**

- **Derivation: Integration by parts**

- **Basis:** $(a+b\,x)^m\,(c+d\,x)^n = (a+b\,x)^{m+n+2}\,\frac{(c+d\,x)^n}{(a+b\,x)^{n+2}}$

- **Rule: If** $m,\,n \in \mathbb{Z} \wedge b\,c-a\,d \neq 0 \wedge 0 < m < -n-2 \wedge 2\,m+n+1 > 0$**, then**

$$\int (a+b\,x)^m\,(c+d\,x)^n\,dx \longrightarrow -\frac{(a+b\,x)^{m+1}\,(c+d\,x)^{n+1}}{(n+1)\,(b\,c-a\,d)} + \frac{b\,(m+n+2)}{(n+1)\,(b\,c-a\,d)}\int (a+b\,x)^m\,(c+d\,x)^{n+1}\,dx$$

- **Program code:**

```
Int[(a_+b_.*x_)^m_.*(c_+d_.*x_)^n_,x_Symbol] :=
  -(a+b*x)^(m+1)*(c+d*x)^(n+1)/((n+1)*(b*c-a*d)) +
  Dist[b*(m+n+2)/((n+1)*(b*c-a*d)),Int[(a+b*x)^m*(c+d*x)^(n+1),x]] /;
FreeQ[{a,b,c,d},x] && IntegersQ[m,n] && NonzeroQ[b*c-a*d] && 0<m<-n-2 && 2*m+n+1>0
```

- **Reference: G&R 2.151, CRC 59b**
- **Derivation: Inverted integration by parts**
- **Rule: If $m, n \in \mathbb{Z} \wedge b c - a d \neq 0 \wedge 0 < n \leq m$, then**

$$\int (a+b x)^m (c+d x)^n \, dx \longrightarrow \frac{(a+b x)^{m+1} (c+d x)^n}{b (m+n+1)} + \frac{n (b c - a d)}{b (m+n+1)} \int (a+b x)^m (c+d x)^{n-1} \, dx$$

- **Program code:**

```
Int[(a_+b_.*x_)^m_*(c_+d_.*x_)^n_.,x_Symbol] :=
  (a+b*x)^(m+1)*(c+d*x)^n/(b*(m+n+1)) +
  Dist[n*(b*c-a*d)/(b*(m+n+1)),Int[(a+b*x)^m*(c+d*x)^(n-1),x]] /;
FreeQ[{a,b,c,d},x] && IntegersQ[m,n] && NonzeroQ[b*c-a*d] && 0<n<=m
```

$$\int x^m (a+b x)^n (c+d x)^p \, dx$$

- **Derivation: Algebraic simplification**
- **Basis: If** $b c + a d = 0$, $(a+b x)(c+d x) = a c + b d x^2$
- **Rule: If** $n \in \mathbb{Z} \wedge b c + a d = 0$, **then**

$$\int x^m (a+b x)^n (c+d x)^n \, dx \longrightarrow \int x^m \left(a c + b d x^2\right)^n dx$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*x_)^n_.*(c_+d_.*x_)^n_.,x_Symbol] :=
  Int[x^m*(a*c+b*d*x^2)^n,x] /;
FreeQ[{a,b,c,d,m},x] && IntegerQ[n] && ZeroQ[b*c+a*d]
```

- **Derivation: Integration by substitution**
- **Basis:** $\partial_x \left(x (a+b x^n)^{p+1}\right) = (a+b x^n)^p (a+b (n (p+1)+1) x^n)$
- **Rule: If** $a d - b c (n (p+1)+1) = 0$, **then**

$$\int (a+b x^n)^p (c+d x^n) \, dx \longrightarrow \frac{c x (a+b x^n)^{p+1}}{a}$$

- **Program code:**

```
Int[(a_+b_.*x_^n_.)^p_.*(c_+d_.*x_^n_.), x_Symbol] :=
  c*x*(a+b*x^n)^(p+1)/a /;
FreeQ[{a,b,c,d,n,p},x] && ZeroQ[a*d-b*c*(n*(p+1)+1)]
```

- **Derivation: Integration by substitution**
- **Basis:** $\partial_x \left(x^{m+1} (a+b x^n)^{p+1}\right) = x^m (a+b x^n)^p (a (m+1) + b (m+n (p+1)+1) x^n)$
- **Rule: If** $m+1 \neq 0 \wedge a d (m+1) - b c (m+n (p+1)+1) = 0$, **then**

$$\int x^m (a+b x^n)^p (c+d x^n) \, dx \longrightarrow \frac{c x^{m+1} (a+b x^n)^{p+1}}{a (m+1)}$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*x_^n_.)^p_.*(c_+d_.*x_^n_.), x_Symbol] :=
  c*x^(m+1)*(a+b*x^n)^(p+1)/(a*(m+1)) /;
FreeQ[{a,b,c,d,m,n,p},x] && NonzeroQ[m+1] && ZeroQ[a*d*(m+1)-b*c*(m+n*(p+1)+1)]
```

- **Derivation: Integration by substitution**
- **Basis:** $\partial_x \left(x^{m+1} (a+b x^n)^{p+1}\right) = x^m (a+b x^n)^p (a (m+1) + b (m+n (p+1)+1) x^n)$
- **Rule: If** $m+q+1 \neq 0 \land a d (m+q+1) - b c (m+q+n (p+1)+1) = 0$**, then**

$$\int x^m (a+b x^n)^p (c x^q + d x^{n+q}) \, dx \longrightarrow \frac{c x^{m+q+1} (a+b x^n)^{p+1}}{a (m+q+1)}$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*x_^n_.)^p_.*(c_.*x_^q_.+d_.*x_^r_.), x_Symbol] :=
  c*x^(m+q+1)*(a+b*x^n)^(p+1)/(a*(m+q+1)) /;
FreeQ[{a,b,c,d,m,n,p,q,r},x] && ZeroQ[r-n-q] && NonzeroQ[m+q+1] &&
    ZeroQ[a*d*(m+q+1)-b*c*(m+q+n*(p+1)+1)]
```

- **Rule: If** $m+n+2 \neq 0 \land f (b c (m+1) + a d (n+1)) - b d e (m+n+2) = 0$**, then**

$$\int (a+b x)^m (c+d x)^n (e+f x) \, dx \longrightarrow \frac{f (a+b x)^{m+1} (c+d x)^{n+1}}{b d (m+n+2)}$$

- **Program code:**

```
Int[(a_+b_.*x_)^m_.*(c_+d_.*x_)^n_.*(e_+f_.*x_), x_Symbol] :=
  f*(a+b*x)^(m+1)*(c+d*x)^(n+1)/(b*d*(m+n+2)) /;
FreeQ[{a,b,c,d,e,f,m,n},x] && NonzeroQ[m+n+2] && ZeroQ[f*(b*c*(m+1)+a*d*(n+1))-b*d*e*(m+n+2)]
```

- **Rule: If** $n, p \in \mathbb{Z} \land 0 < n \leq 2 \land p > 5$**, then**

$$\int x (a+b x)^n (c+d x)^p \, dx \longrightarrow \frac{(a+b x)^{n+1} (c+d x)^{p+1}}{b d (2+n+p)} - \frac{b c (n+1) + a d (p+1)}{b d (2+n+p)} \int (a+b x)^n (c+d x)^p \, dx$$

- **Program code:**

```
Int[x_*(a_+b_.*x_)^n_.*(c_+d_.*x_)^p_.,x_Symbol] :=
  (a+b*x)^(n+1)*(c+d*x)^(p+1)/(b*d*(2+n+p)) -
  Dist[(b*c*(n+1)+a*d*(p+1))/(b*d*(2+n+p)), Int[(a+b*x)^n*(c+d*x)^p, x]] /;
FreeQ[{a,b,c,d,n,p},x] && IntegersQ[n,p] && 0<n<=2 && p>5
```

- **Rule: If** $m, n, p \in \mathbb{Z} \land 0 < m \leq 2 \land 0 < n \leq 2 \land p > 5$, **then**

$$\int x^m (a+b x)^n (c+d x)^p \, dx \longrightarrow \frac{x^{m-1} (a+b x)^{n+1} (c+d x)^{p+1}}{b d (1+m+n+p)} - \frac{a c (m-1)}{b d (1+m+n+p)} \int x^{m-2} (a+b x)^n (c+d x)^p \, dx - \frac{b c (m+n) + a d (m+p)}{b d (1+m+n+p)} \int x^{m-1} (a+b x)^n (c+d x)^p \, dx$$

- **Program code:**

```
Int[x_^m_*(a_+b_.*x_)^n_.*(c_+d_.*x_)^p_.,x_Symbol] :=
  x^(m-1)*(a+b*x)^(n+1)*(c+d*x)^(p+1)/(b*d*(1+m+n+p)) -
  Dist[a*c*(m-1)/(b*d*(1+m+n+p)), Int[x^(m-2)*(a+b*x)^n*(c+d*x)^p, x]] -
  Dist[(b*c*(m+n)+a*d*(m+p))/(b*d*(1+m+n+p)), Int[x^(m-1)*(a+b*x)^n*(c+d*x)^p, x]] /;
FreeQ[{a,b,c,d,n,p},x] && IntegersQ[m,n,p] && 0<m<=2 && 0<n<=2 && p>5
```