Integration Rules for

$$\int \left(\sin^j(z)\right)^m \left(a + b\sin^k(z)\right)^n \, dz \text{ when } j^2 = 1 \bigwedge k^2 = 1 \bigwedge a^2 = b^2$$

Domain Map



Legend:

- The rule number in a colored region indicates the rule to use for integrals in that region.
- The rule number next to a colored line indicates the rule to use for integrals on that line.
- A white region or line indicates there is no rule for integrals in that region or on that line.
- A solid black line indicates integrals on that line are handled by rules in another section.
- A dashed black line on the border of a region indicates integrals on that border are handled by the rule for that region.
- The arrow(s) following a rule number indicates the direction the rule drives integrands in the n×m exponent plane.
- A ⋄ following a rule number indicates the rule transforms the integrand into a form handled by another section.
- A red (stop) disk indicates the terminal rule to use for the point at the center of the disk.
- A cyan disk indicates the non-terminal rule to use for the point at the center of the disk.

Integration Rules for

$\int (a + b \sin^k(z))^n \, dz$ when $k^2 = 1 \bigwedge a^2 = b^2$

Rule a: $\int \frac{1}{a + b \operatorname{Sin}[c + d x]^k} \, dx$

- **Reference: G&R 2.555.3', CRC 337', A&S 4.3.134'/5'**
- **Derivation: Rule 1b with** $m = 0$, $k = 1$ **and** $n = -1$
- **Rule a1: If** $a^2 - b^2 = 0$, **then**

$$\int \frac{1}{a + b \operatorname{Sin}[c + d x]} \, dx \longrightarrow -\frac{\operatorname{Cos}[c + d x]}{d (b + a \operatorname{Sin}[c + d x])}$$

- **Program code:**

```
Int[1/(a_+b_.*sin[c_.+d_.*x_]),x_Symbol] :=
  -Cos[c+d*x]/(d*(b+a*Sin[c+d*x])) /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2-b^2]
```

- **Reference: G&R 2.555.4', CRC 338'/9', A&S 4.3.134'/5'**

```
Int[1/(a_+b_.*Cos[c_.+d_.*x_]),x_Symbol] :=
  Sin[c+d*x]/(d*(b+a*Cos[c+d*x])) /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2-b^2]
```

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{1}{a+b z} = \frac{1}{a} - \frac{b z}{a (a+b z)}$
- **Note: The rule for integrands of the same form when** $a^2 - b^2 \neq 0$ **could subsume this rule, but the resulting antiderivative will look less like the integrand involving sines instead of cosecants.**
- **Rule a2: If** $a^2 - b^2 = 0$, **then**

$$\int \frac{1}{a + b \operatorname{Csc}[c + d x]} \, dx \longrightarrow \frac{x}{a} - \frac{b}{a} \int \frac{\operatorname{Csc}[c + d x]}{a + b \operatorname{Csc}[c + d x]} \, dx$$

- **Program code:**

```
Int[1/(a_+b_.*sin[c_.+d_.*x_]^(-1)),x_Symbol] :=
  x/a - Dist[b/a,Int[1/(sin[c+d*x]*(a+b/sin[c+d*x])),x]] /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2-b^2]
```

Rule b: $\int \sqrt{a + b\,\mathrm{Sin}[c + d\,x]^k}\,dx$

- **Derivation: Rule 3b with $m = 0$, $k = 1$ and $n = \frac{1}{2}$**
- **Rule b1: If $a^2 - b^2 = 0$, then**

$$\int \sqrt{a + b\,\mathrm{Sin}[c + d\,x]}\,dx \longrightarrow -\frac{2\,b\,\mathrm{Cos}[c + d\,x]}{d\,\sqrt{a + b\,\mathrm{Sin}[c + d\,x]}}$$

- **Program code:**

```
Int[Sqrt[a_+b_.*sin[c_.+d_.*x_]],x_Symbol] :=
  -2*b*Cos[c+d*x]/(d*Sqrt[a+b*Sin[c+d*x]]) /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2-b^2]
```

```
Int[Sqrt[a_+b_.*Cos[c_.+d_.*x_]],x_Symbol] :=
  2*b*Sin[c+d*x]/(d*Sqrt[a+b*Cos[c+d*x]]) /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2-b^2]
```

- **Author: Martin on sci.math.symbolic on 10 March 2011**
- **Rule b2: If $a^2 - b^2 = 0$, then**

$$\int \sqrt{a + b\,\mathrm{Csc}[c + d\,x]}\,dx \longrightarrow -\frac{2\,\sqrt{a}}{d}\,\mathrm{ArcTan}\left[\frac{\sqrt{a}\,\mathrm{Cot}[c + d\,x]}{\sqrt{a + b\,\mathrm{Csc}[c + d\,x]}}\right]$$

- **Program code:**

```
Int[Sqrt[a_+b_.*sin[c_.+d_.*x_]^(-1)],x_Symbol] :=
  -2*Sqrt[a]/d*ArcTan[(Sqrt[a]*Cot[c+d*x])/Sqrt[a+b*Csc[c+d*x]]] /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2-b^2]
```

Rule c: $\int \frac{1}{\sqrt{a + b \operatorname{Sin}[c + d x]}} \, dx$

- **Note: Although not essential, this rule produces a simpler antiderivative than rule c3.**

- **Rule c1: If $a - b = 0$, then**

$$\int \frac{1}{\sqrt{a + b \operatorname{Cos}[c + d x]}} \, dx \longrightarrow \frac{2}{d \sqrt{a + b \operatorname{Cos}[c + d x]}} \operatorname{Cos}\left[\frac{c + d x}{2}\right] \operatorname{ArcTanh}\left[\operatorname{Sin}\left[\frac{c + d x}{2}\right]\right]$$

- **Program code:**

```
Int[1/Sqrt[a_+b_.*sin[c_.+Pi/2+d_.*x_]],x_Symbol] :=
  2/(d*Sqrt[a+b*Cos[c+d*x]])*Cos[(c+d*x)/2]*ArcTanh[Sin[(c+d*x)/2]] /;
FreeQ[{a,b,c,d},x] && ZeroQ[a-b]
```

- **Note: Although not essential, this rule produces a simpler antiderivative than rule c3.**

- **Rule c2: If $a + b = 0$, then**

$$\int \frac{1}{\sqrt{a + b \operatorname{Cos}[c + d x]}} \, dx \longrightarrow -\frac{2}{d \sqrt{a + b \operatorname{Cos}[c + d x]}} \operatorname{Sin}\left[\frac{c + d x}{2}\right] \operatorname{ArcTanh}\left[\operatorname{Cos}\left[\frac{c + d x}{2}\right]\right]$$

- **Program code:**

```
Int[1/Sqrt[a_+b_.*sin[c_.+Pi/2+d_.*x_]],x_Symbol] :=
  -2/(d*Sqrt[a+b*Cos[c+d*x]])*Sin[(c+d*x)/2]*ArcTanh[Cos[(c+d*x)/2]] /;
FreeQ[{a,b,c,d},x] && ZeroQ[a+b]
```

- **Rule c3: If $a^2 - b^2 = 0$, then**

$$\int \frac{1}{\sqrt{a + b \operatorname{Sin}[c + d x]}} \, dx \longrightarrow \frac{2}{d \sqrt{a + b \operatorname{Sin}[c + d x]}} \operatorname{Cos}\left[\frac{c + d x}{2} - \frac{\pi b}{4 a}\right] \operatorname{ArcTanh}\left[\operatorname{Sin}\left[\frac{c + d x}{2} - \frac{\pi b}{4 a}\right]\right]$$

- **Program code:**

```
Int[1/Sqrt[a_+b_.*sin[c_.+d_.*x_]],x_Symbol] :=
  2/(d*Sqrt[a+b*Sin[c+d*x]])*Cos[(c+d*x)/2-Pi*b/(4*a)]*ArcTanh[Sin[(c+d*x)/2-Pi*b/(4*a)]] /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2-b^2]
```

Rules 15 – 16: $\int (a + b \operatorname{Csc}[c + d x])^n \, dx$

- **Derivation:** Rule 6 with $m = 0$ and $k = -1$
- **Rule 15:** If $a^2 - b^2 = 0 \land n < -1$, then

$$\int (a + b \operatorname{Csc}[c + d x])^n \, dx \longrightarrow -\frac{\operatorname{Cot}[c + d x]\,(a + b \operatorname{Csc}[c + d x])^n}{d\,(2 n + 1)} + \frac{1}{a^2\,(2 n + 1)} \int (a\,(2 n + 1) - b\,(n + 1) \operatorname{Csc}[c + d x])\,(a + b \operatorname{Csc}[c + d x])^{n+1} \, dx$$

- **Program code:**

```
Int[(a_+b_.*sin[c_.+d_.*x_]^(-1))^n_,x_Symbol] :=
  -Cot[c+d*x]*(a+b*Csc[c+d*x])^n/(d*(2*n+1)) +
  Dist[1/(a^2*(2*n+1)),Int[(a*(2*n+1)-(b*(n+1))*sin[c+d*x]^(-1))*(a+b*sin[c+d*x]^(-1))^(n+1),x]] /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2-b^2] && RationalQ[n] && n<-1
```

- **Derivation:** Rule 3a with $m = 0$ and $k = -1$
- **Rule 16:** If $a^2 - b^2 = 0 \land n > 1 \land n \neq 2$, then

$$\int (a + b \operatorname{Csc}[c + d x])^n \, dx \longrightarrow -\frac{b^2 \operatorname{Cot}[c + d x]\,(a + b \operatorname{Csc}[c + d x])^{n-2}}{d\,(n - 1)} + \frac{a}{n - 1} \int (a\,(n - 1) + b\,(3 n - 4) \operatorname{Csc}[c + d x])\,(a + b \operatorname{Csc}[c + d x])^{n-2} \, dx$$

- **Program code:**

```
Int[(a_+b_.*sin[c_.+d_.*x_]^(-1))^n_,x_Symbol] :=
  -b^2*Cot[c+d*x]*(a+b*Csc[c+d*x])^(n-2)/(d*(n-1)) +
  Dist[a/(n-1),Int[(a*(n-1)+(b*(3*n-4))*sin[c+d*x]^(-1))*(a+b*sin[c+d*x]^(-1))^(n-2),x]] /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2-b^2] && RationalQ[n] && n>1 && n≠2
```

Integration Rules for

$\int\left(\sin^j(z)\right)^m\left(a+b\sin^k(z)\right)^n\,dz$ when $j^2=1\bigwedge k^2=1\bigwedge a^2=b^2$

Rule d: $\int\frac{\sqrt{a+b\,Sin[c+d\,x]}}{Sin[c+d\,x]}\,dx$

- **Derivation: Piecewise constant extraction and trig substitution**
- **Basis: If** $a-b=0$, **then** $\partial_z\frac{Cos\left[\frac{z}{2}\right]}{\sqrt{a+b\,Cos[z]}}=0$
- **Basis: If** $a-b=0$, **then** $a+b\,Cos[z]=2\,a\,Cos\left[\frac{z}{2}\right]^2$
- **Note: Although not essential, this rule produces a simpler antiderivative than rule d3.**
- **Rule d1: If** $a-b=0$, **then**

$$\int\frac{\sqrt{a+b\,Cos[c+d\,x]}}{Cos[c+d\,x]}\,dx\;\rightarrow\;\frac{2\,a\,Cos\left[\frac{c+d\,x}{2}\right]}{\sqrt{a+b\,Cos[c+d\,x]}}\int\frac{Cos\left[\frac{c+d\,x}{2}\right]}{Cos[c+d\,x]}\,dx$$

$$\rightarrow\;\frac{2\sqrt{2}\,b\,Cos\left[\frac{c+d\,x}{2}\right]}{d\sqrt{a+b\,Cos[c+d\,x]}}\,ArcTanh\left[\sqrt{2}\,Sin\left[\frac{c+d\,x}{2}\right]\right]$$

- **Program code:**

```
Int[Sqrt[a_+b_.*sin[c_.+Pi/2+d_.*x_]]/sin[c_.+Pi/2+d_.*x_],x_Symbol] :=
  2*Sqrt[2]*b*Cos[(c+d*x)/2]/(d*Sqrt[a+b*Cos[c+d*x]])*ArcTanh[Sqrt[2]*Sin[(c+d*x)/2]] /;
FreeQ[{a,b,c,d},x] && ZeroQ[a-b]
```

- **Derivation: Piecewise constant extraction and trig substitution**
- **Basis: If $a+b=0$, then $\partial_z \frac{\text{Sin}\left[\frac{z}{2}\right]}{\sqrt{a+b\,\text{Cos}[z]}} = 0$**
- **Basis: If $a+b=0$, then $a+b\,\text{Cos}[z] = 2\,a\,\text{Sin}\left[\frac{z}{2}\right]^2$**
- **Note: Although not essential, this rule produces a simpler antiderivative than rule d3.**
- **Rule d2: If $a+b=0$, then**

$$\int \frac{\sqrt{a+b\,\text{Cos}[c+d\,x]}}{\text{Cos}[c+d\,x]}\,dx \longrightarrow \frac{2\,a\,\text{Sin}\left[\frac{c+d\,x}{2}\right]}{\sqrt{a+b\,\text{Cos}[c+d\,x]}}\int \frac{\text{Sin}\left[\frac{c+d\,x}{2}\right]}{\text{Cos}[c+d\,x]}\,dx$$

$$\longrightarrow \frac{2\sqrt{2}\,a\,\text{Sin}\left[\frac{c+d\,x}{2}\right]}{d\sqrt{a+b\,\text{Cos}[c+d\,x]}}\,\text{ArcTanh}\left[\sqrt{2}\,\text{Cos}\left[\frac{c+d\,x}{2}\right]\right]$$

- **Program code:**

```
Int[Sqrt[a_+b_.*sin[c_.+Pi/2+d_.*x_]]/sin[c_.+Pi/2+d_.*x_],x_Symbol] :=
  2*Sqrt[2]*a*Sin[(c+d*x)/2]/(d*Sqrt[a+b*Cos[c+d*x]])*ArcTanh[Sqrt[2]*Cos[(c+d*x)/2]] /;
FreeQ[{a,b,c,d},x] && ZeroQ[a+b]
```

- **Derivation: Piecewise constant extraction and trig substitution**
- **Rule d3: If $a^2-b^2=0$, then**

$$\int \frac{\sqrt{a+b\,\text{Sin}[c+d\,x]}}{\text{Sin}[c+d\,x]}\,dx \longrightarrow \frac{2\sqrt{2}\,b\,\text{Cos}\left[\frac{c+d\,x}{2}-\frac{\pi\,b}{4\,a}\right]}{d\sqrt{a+b\,\text{Sin}[c+d\,x]}}\,\text{ArcTanh}\left[\sqrt{2}\,\text{Sin}\left[\frac{c+d\,x}{2}-\frac{\pi\,b}{4\,a}\right]\right]$$

- **Program code:**

```
Int[Sqrt[a_+b_.*sin[c_.+d_.*x_]]/sin[c_.+d_.*x_],x_Symbol] :=
  2*Sqrt[2]*b*Cos[(c+d*x)/2-Pi*b/(4*a)]/(d*Sqrt[a+b*Sin[c+d*x]])*
    ArcTanh[Sqrt[2]*Sin[(c+d*x)/2-Pi*b/(4*a)]] /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2-b^2]
```

Rule e: $\int \frac{1}{\operatorname{Sin}[c+d\,x]^{\frac{k+1}{2}} \sqrt{a+b\operatorname{Sin}[c+d\,x]^k}}\,dx$

- **Author: Martin on sci.math.symbolic on 10 March 2011**
- **Derivation: Algebraic expansion**
- **Basis: If** $k^2 = 1$, **then** $\frac{1}{z^{\frac{k+1}{2}} \sqrt{a+b z^k}} = \frac{\sqrt{a+b z^k}}{a z^{\frac{k+1}{2}}} - \frac{b z^{\frac{k-1}{2}}}{a \sqrt{a+b z^k}}$
- **Rule e: If** $k^2 = 1 \land a^2 - b^2 = 0$, **then**

$$\int \frac{1}{\operatorname{Sin}[c+d\,x]^{\frac{k+1}{2}} \sqrt{a+b\operatorname{Sin}[c+d\,x]^k}}\,dx \longrightarrow \frac{1}{a}\int \frac{\sqrt{a+b\operatorname{Sin}[c+d\,x]^k}}{\operatorname{Sin}[c+d\,x]^{\frac{k+1}{2}}}\,dx - \frac{b}{a}\int \frac{\operatorname{Sin}[c+d\,x]^{\frac{k-1}{2}}}{\sqrt{a+b\operatorname{Sin}[c+d\,x]^k}}\,dx$$

- **Program code:**

```
Int[1/(sin[c_.+d_.*x_]*Sqrt[a_+b_.*sin[c_.+d_.*x_]]),x_Symbol] :=
  Dist[1/a,Int[Sqrt[a+b*sin[c+d*x]]/sin[c+d*x],x]] -
  Dist[b/a,Int[1/Sqrt[a+b*sin[c+d*x]],x]] /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2-b^2]
```

```
Int[1/Sqrt[a_+b_.*sin[c_.+d_.*x_]^(-1)],x_Symbol] :=
  1/a*Int[Sqrt[a+b*sin[c+d*x]^(-1)],x] -
  b/a*Int[sin[c+d*x]^(-1)/Sqrt[a+b*sin[c+d*x]^(-1)],x] /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2-b^2]
```

Rule f: $\int \frac{1}{\sqrt{\text{Sin}[c+d\,x]}\,\sqrt{a+b\,\text{Sin}[c+d\,x]}}\,dx$

- **Basis:** $F(z\mid 0)=z$
- **Note: This is a special case of the rule for** $a^2 \neq b^2$.
- **Rule f1: If** $a-b=0\wedge a>0$, **then**

$$\int \frac{1}{\sqrt{\text{Sin}[c+d\,x]}\,\sqrt{a+b\,\text{Sin}[c+d\,x]}}\,dx \to \frac{\sqrt{2}}{d\,\sqrt{a}}\,\text{ArcSin}\left[\text{Tan}\left[\frac{c+d\,x}{2}-\frac{\pi}{4}\right]\right]$$

- **Program code:**

```
Int[1/(Sqrt[sin[c_.+Pi/2+d_.*x_]]*Sqrt[a_+b_.*sin[c_.+Pi/2+d_.*x_]]),x_Symbol] :=
  Sqrt[2]/(d*Sqrt[a])*ArcSin[Tan[(c+d*x)/2]] /;
FreeQ[{a,b,c,d},x] && ZeroQ[a-b] && PositiveQ[a]
```

```
Int[1/(Sqrt[sin[c_.+d_.*x_]]*Sqrt[a_+b_.*sin[c_.+d_.*x_]]),x_Symbol] :=
  Sqrt[2]/(d*Sqrt[a])*ArcSin[Tan[(c+d*x)/2-Pi/4]] /;
FreeQ[{a,b,c,d},x] && ZeroQ[a-b] && PositiveQ[a]
```

- **Author: Martin 10 March 2011**
- **Derivation: ???**
- **Rule f2: If** $a^2-b^2=0$, **then**

$$\int \frac{1}{\sqrt{\text{Sin}[c+d\,x]}\,\sqrt{a+b\,\text{Sin}[c+d\,x]}}\,dx \to -\frac{\sqrt{2}\,\sqrt{b}}{a\,d}\,\text{ArcTan}\left[\frac{\sqrt{b}\,\text{Cos}[c+d\,x]}{\sqrt{2}\,\sqrt{\text{Sin}[c+d\,x]}\,\sqrt{a+b\,\text{Sin}[c+d\,x]}}\right]$$

- **Program code:**

```
Int[1/(Sqrt[sin[c_.+d_.*x_]]*Sqrt[a_+b_.*sin[c_.+d_.*x_]]),x_Symbol] :=
  -Sqrt[2]*Sqrt[b]/(a*d)*ArcTan[Sqrt[b]*Cos[c+d*x]/(Sqrt[2]*Sqrt[Sin[c+d*x]]*Sqrt[a+b*Sin[c+d*x]])]
FreeQ[{a,b,c,d},x] && ZeroQ[a^2-b^2] && Not[ZeroQ[a-b] && PositiveQ[a]]
```

Rule g: $\int \frac{\sqrt{a + b \operatorname{Sin}[c + d x]}}{\sqrt{\operatorname{Sin}[c + d x]}} \, dx$

- **Author: Martin 10 March 2011**
- **Derivation: ???**
- **Rule g: If $a^2 - b^2 = 0$, then**

$$\int \frac{\sqrt{a + b \operatorname{Sin}[c + d x]}}{\sqrt{\operatorname{Sin}[c + d x]}} \, dx \longrightarrow -\frac{2\sqrt{b}}{d} \operatorname{ArcTan}\left[\frac{\sqrt{b} \operatorname{Cos}[c + d x]}{\sqrt{\operatorname{Sin}[c + d x]} \sqrt{a + b \operatorname{Sin}[c + d x]}}\right]$$

- **Program code:**

```
Int[Sqrt[a_+b_.*sin[c_.+d_.*x_]]/Sqrt[sin[c_.+d_.*x_]],x_Symbol] :=
  -2*Sqrt[b]/d*ArcTan[Sqrt[b]*Cos[c+d*x]/(Sqrt[Sin[c+d*x]]*Sqrt[a+b*Sin[c+d*x]])] /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2-b^2]
```

$$\text{Rule h:}\quad \int \frac{\sqrt{\text{Sin}[c+d\,x]}}{\sqrt{a+b\,\text{Sin}[c+d\,x]}}\,dx$$

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{\sqrt{z}}{\sqrt{a+b\,z}} = \frac{\sqrt{a+b\,z}}{b\,\sqrt{z}} - \frac{a}{b\,\sqrt{z}\,\sqrt{a+b\,z}}$
- **Rule h: If** $a^2-b^2=0$**, then**

$$\int \frac{\sqrt{\text{Sin}[c+d\,x]}}{\sqrt{a+b\,\text{Sin}[c+d\,x]}}\,dx \longrightarrow \frac{1}{b}\int \frac{\sqrt{a+b\,\text{Sin}[c+d\,x]}}{\sqrt{\text{Sin}[c+d\,x]}}\,dx - \frac{a}{b}\int \frac{1}{\sqrt{\text{Sin}[c+d\,x]}\,\sqrt{a+b\,\text{Sin}[c+d\,x]}}\,dx$$

- **Program code:**

```
Int[Sqrt[sin[c_.+d_.*x_]]/Sqrt[a_+b_.*sin[c_.+d_.*x_]],x_Symbol] :=
  Dist[1/b,Int[Sqrt[a+b*sin[c+d*x]]/Sqrt[sin[c+d*x]],x]] -
  Dist[a/b,Int[1/(Sqrt[sin[c+d*x]]*Sqrt[a+b*sin[c+d*x]]),x]] /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2-b^2]
```

$$\text{Rule i:}\quad \int \frac{\left(\text{Sin}[c+d\,x]^j\right)^{m/2}}{\left(a+b\,\text{Sin}[c+d\,x]^k\right)^2}\,dx$$

- **Derivation: Rule 1b with $n=-2$ and $2\,j\,k\,m+k-2=0$**
- **Rule i: If $j^2=k^2=1\land a^2-b^2=0\land 2\,j\,k\,m+k-2=0$, then**

$$\int \frac{\left(\text{Sin}[c+d\,x]^j\right)^m}{\left(a+b\,\text{Sin}[c+d\,x]^k\right)^2}\,dx \longrightarrow -\frac{a\,\text{Cos}[c+d\,x]\,\left(\text{Sin}[c+d\,x]^j\right)^m}{3\,b\,d\,\left(a+b\,\text{Sin}[c+d\,x]^k\right)^2}+\frac{1}{6\,a\,b}\int \left(\text{Sin}[c+d\,x]^j\right)^{m-j\,k}\,dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_./(a_+b_.*sin[c_.+d_.*x_]^k_.)^2,x_Symbol] :=
  -a*Cos[c+d*x]*(Sin[c+d*x]^j)^m/(3*b*d*(a+b*Sin[c+d*x]^k)^2) +
  1/(6*a*b)*Int[(sin[c+d*x]^j)^(m-j*k),x] /;
FreeQ[{a,b,c,d},x] && OneQ[j^2,k^2] && ZeroQ[a^2-b^2] && RationalQ[m] &&
  ZeroQ[2*j*k*m+k-2]
```

- **Derivation: Rule 1b with $2\,j\,k\,m+n+k=0$**
- **Note: Unfortunately this interesting looking rule seems to be of no use except for the above special case when $n=-2$.**
- **Rule: If $j^2=k^2=1\land a^2-b^2=0\land 2\,j\,k\,m+n+k=0\land j\,k\,m>0\land n<-1$, then**

$$\int \left(\text{Sin}[c+d\,x]^j\right)^m\,\left(a+b\,\text{Sin}[c+d\,x]^k\right)^n\,dx \longrightarrow$$
$$\frac{a\,\text{Cos}[c+d\,x]\,\left(\text{Sin}[c+d\,x]^j\right)^m\,\left(a+b\,\text{Sin}[c+d\,x]^k\right)^n}{b\,d\,(2\,n+1)}+$$
$$\frac{n+1}{2\,a\,b\,(2\,n+1)}\int \left(\text{Sin}[c+d\,x]^j\right)^{m-j\,k}\,\left(a+b\,\text{Sin}[c+d\,x]^k\right)^{n+2}\,dx$$

- **Program code:**

```
(* Int[(sin[c_.+d_.*x_]^j_.)^m_.*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  a*Cos[c+d*x]*(Sin[c+d*x]^j)^m*(a+b*Sin[c+d*x]^k)^n/(b*d*(2*n+1)) +
  Dist[(n+1)/(2*a*b*(2*n+1)),Int[(sin[c+d*x]^j)^(m-j*k)*(a+b*sin[c+d*x]^k)^(n+2),x]] /;
FreeQ[{a,b,c,d},x] && OneQ[j^2,k^2] && ZeroQ[a^2-b^2] && RationalQ[m,n] &&
  ZeroQ[2*j*k*m+n+k] && j*k*m>0 && n<-1 *)
```

$$\int \operatorname{Csc}[c+d\,x]\,\left(a+b\operatorname{Csc}[c+d\,x]\right)^n\,dx$$

- **Note: Although the integrand equals $\frac{1}{b+a\sin[c+d\,x]}$ which is easily integrated, this antiderivative is more similar in form to the integrand.**
- **Rule: If $a^2-b^2=0$, then**

$$\int \frac{\operatorname{Csc}[c+d\,x]}{a+b\operatorname{Csc}[c+d\,x]}\,dx \longrightarrow -\frac{\operatorname{Cot}[c+d\,x]}{d\,(b+a\operatorname{Csc}[c+d\,x])}$$

- **Program code:**

```
Int[sin[c_.+d_.*x_]^(-1)/(a_+b_.*sin[c_.+d_.*x_]^(-1)),x_Symbol] :=
  -Cot[c+d*x]/(d*(b+a*Csc[c+d*x])) /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2-b^2]
```

- **Derivation: Rule 3b with $j\,m=-1$, $k=-1$ and $n=\frac{1}{2}$**
- **Rule: If $a^2-b^2=0$, then**

$$\int \operatorname{Csc}[c+d\,x]\,\sqrt{a+b\operatorname{Csc}[c+d\,x]}\,dx \longrightarrow -\frac{2\,b\operatorname{Cot}[c+d\,x]}{d\sqrt{a+b\operatorname{Csc}[c+d\,x]}}$$

- **Program code:**

```
Int[sin[c_.+d_.*x_]^(-1)*Sqrt[a_+b_.*sin[c_.+d_.*x_]^(-1)],x_Symbol] :=
  -2*b*Cot[c+d*x]/(d*Sqrt[a+b*Csc[c+d*x]]) /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2-b^2]
```

- **Author: Martin on sci.math.symbolic on 10 March 2011**
- **Rule: If $a^2-b^2=0$, then**

$$\int \frac{\operatorname{Csc}[c+d\,x]}{\sqrt{a+b\operatorname{Csc}[c+d\,x]}}\,dx \longrightarrow -\frac{\sqrt{2\,a}}{b\,d}\operatorname{ArcTan}\left[\frac{\sqrt{a}\operatorname{Cot}[c+d\,x]}{\sqrt{2}\sqrt{a+b\operatorname{Csc}[c+d\,x]}}\right]$$

- **Program code:**

```
Int[1/(sin[c_.+d_.*x_]*Sqrt[a_+b_.*sin[c_.+d_.*x_]^(-1)]),x_Symbol] :=
  -Sqrt[2*a]/(b*d)*ArcTan[Sqrt[a]*Cot[c + d*x]/(Sqrt[2]*Sqrt[a + b*Csc[c + d*x]])] /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2-b^2]
```

$$\int \mathrm{Csc}[c+d\,x]^2\,(a+b\,\mathrm{Csc}[c+d\,x])^n\,dx$$

- **Derivation: Rule 1a with** $j\,m = -2$ **and** $k = -1$
- **Rule: If** $a^2-b^2=0\,\wedge\,n\neq -1\,\wedge\,n\neq 1\,\wedge\,n\neq 2$, **then**

$$\int \mathrm{Csc}[c+d\,x]^2\,(a+b\,\mathrm{Csc}[c+d\,x])^n\,dx\,\longrightarrow$$
$$-\frac{\mathrm{Cot}[c+d\,x]\,(a+b\,\mathrm{Csc}[c+d\,x])^n}{d\,(2\,n+1)}+\frac{n}{b\,(2\,n+1)}\int \mathrm{Csc}[c+d\,x]\,(a+b\,\mathrm{Csc}[c+d\,x])^{n+1}\,dx$$

- **Program code:**

```
Int[sin[c_.+d_.*x_]^(-2)*(a_+b_.*sin[c_.+d_.*x_]^(-1))^n_.,x_Symbol] :=
  -Cot[c+d*x]*(a+b*Csc[c+d*x])^n/(d*(2*n+1)) +
  Dist[n/(b*(2*n+1)),Int[sin[c+d*x]^(-1)*(a+b*sin[c+d*x]^(-1))^(n+1),x]] /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2-b^2] && RationalQ[n] && n<-1
```

- **Derivation: Rule 2 with** $j\,m = -2$ **and** $k = -1$
- **Rule: If** $a^2-b^2=0\,\wedge\,n>-1\,\wedge\,n\neq 1\,\wedge\,n\neq 2$, **then**

$$\int \mathrm{Csc}[c+d\,x]^2\,(a+b\,\mathrm{Csc}[c+d\,x])^n\,dx\,\longrightarrow$$
$$-\frac{\mathrm{Cot}[c+d\,x]\,(a+b\,\mathrm{Csc}[c+d\,x])^n}{d\,(n+1)}+\frac{b\,n}{a\,(n+1)}\int \mathrm{Csc}[c+d\,x]\,(a+b\,\mathrm{Csc}[c+d\,x])^{n}\,dx$$

- **Program code:**

```
Int[sin[c_.+d_.*x_]^(-2)*(a_+b_.*sin[c_.+d_.*x_]^(-1))^n_.,x_Symbol] :=
  -Cot[c+d*x]*(a+b*Csc[c+d*x])^n/(d*(n+1)) +
  Dist[b*n/(a*(n+1)),Int[sin[c+d*x]^(-1)*(a+b*sin[c+d*x]^(-1))^n,x]] /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2-b^2] && RationalQ[n] && n>-1 && n≠1 && n≠2
```

$$\int \left(\text{Sin}[c+d\,x]^j\right)^{m/2} (a+b\,\text{Csc}[c+d\,x])^{n/2}\,dx$$

- **Derivation: Algebraic expansion**

- **Basis:** $\frac{1}{\sqrt{a+b/z}} = \frac{z\sqrt{a+b/z}}{b} - \frac{a\,z}{b\sqrt{a+b/z}}$

- **Rule: If** $j^2 = 1 \land a^2 - b^2 = 0 \land j\,m = -3$**, then**

$$\int \frac{\left(\text{Sin}[c+d\,x]^j\right)^{m/2}}{\sqrt{a+b\,\text{Csc}[c+d\,x]}}\,dx \longrightarrow \frac{1}{b}\int \left(\text{Sin}[c+d\,x]^j\right)^{m/2+j}\sqrt{a+b\,\text{Csc}[c+d\,x]}\,dx - \frac{a}{b}\int \frac{\left(\text{Sin}[c+d\,x]^j\right)^{m/2+j}}{\sqrt{a+b\,\text{Csc}[c+d\,x]}}\,dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_/Sqrt[a_+b_.*sin[c_.+d_.*x_]^(-1)],x_Symbol] :=
  Dist[1/b,Int[(sin[c+d*x]^j)^(m+j)*Sqrt[a+b*sin[c+d*x]^(-1)],x]] -
  Dist[a/b,Int[(sin[c+d*x]^j)^(m+j)/Sqrt[a+b*sin[c+d*x]^(-1)],x]] /;
FreeQ[{a,b,c,d},x] && OneQ[j^2] && ZeroQ[a^2-b^2] && RationalQ[m] && j*m==-3/2
```

- **Derivation: Rule 5 with** $j\,m = \frac{1}{2}$, $k = -1$ **and** $n = -\frac{1}{2}$

- **Rule: If** $j^2 = 1 \land a^2 - b^2 = 0 \land j\,m = 1$**, then**

$$\int \frac{\left(\text{Sin}[c+d\,x]^j\right)^{m/2}}{\sqrt{a+b\,\text{Csc}[c+d\,x]}}\,dx \longrightarrow$$

$$-\frac{2\,\text{Cos}[c+d\,x]}{d\left(\text{Sin}[c+d\,x]^j\right)^{m/2}\sqrt{a+b\,\text{Csc}[c+d\,x]}} - \frac{a}{b}\int \frac{1}{\left(\text{Sin}[c+d\,x]^j\right)^{m/2}\sqrt{a+b\,\text{Csc}[c+d\,x]}}\,dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_/Sqrt[a_+b_.*sin[c_.+d_.*x_]^(-1)],x_Symbol] :=
  -2*Cos[c+d*x]/(d*(Sin[c+d*x]^j)^m*Sqrt[a+b*Csc[c+d*x]]) -
  Dist[a/b,Int[1/((sin[c+d*x]^j)^m*Sqrt[a+b*sin[c+d*x]^(-1)]),x]] /;
FreeQ[{a,b,c,d},x] && OneQ[j^2] && ZeroQ[a^2-b^2] && RationalQ[m] && j*m==1/2
```

- **Derivation:** Rule 4a with $j\,m = \frac{1}{2}$, $k = -1$ and $n = \frac{3}{2}$
- **Rule:** If $j^2 = 1 \land a^2 - b^2 = 0 \land j\,m = 1$, then

$$\int \left(\text{Sin}[c+d\,x]^j\right)^{m/2} (a+b\,\text{Csc}[c+d\,x])^{3/2}\,dx \longrightarrow$$

$$-\frac{2\,a^2\,\text{Cos}[c+d\,x]}{d\left(\text{Sin}[c+d\,x]^j\right)^{m/2}\sqrt{a+b\,\text{Csc}[c+d\,x]}}+b\int\frac{\sqrt{a+b\,\text{Csc}[c+d\,x]}}{\left(\text{Sin}[c+d\,x]^j\right)^{m/2}}\,dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_*(a_+b_.*sin[c_.+d_.*x_]^(-1))^(3/2),x_Symbol] :=
  -2*a^2*Cos[c+d*x]/(d*(Sin[c+d*x]^j)^m*Sqrt[a+b*Csc[c+d*x]]) +
  Dist[b,Int[Sqrt[a+b*sin[c+d*x]^(-1)]/(sin[c+d*x]^j)^m,x]] /;
FreeQ[{a,b,c,d},x] && OneQ[j^2] && ZeroQ[a^2-b^2] && RationalQ[m] && j*m==1/2
```

- **Derivation:** Piecewise constant extraction
- **Basis:** If $\partial_z \frac{\sqrt{b+a\,f[z]}}{\sqrt{f[z]}\,\sqrt{a+b\,f[z]^{-1}}} = 0$
- **Rule:** If $j^2 = 1 \land a^2 - b^2 = 0 \land j\,m = -1 \land n^2 = 1$, then

$$\int \left(\text{Sin}[c+d\,x]^j\right)^{m/2} (a+b\,\text{Csc}[c+d\,x])^{n/2}\,dx \longrightarrow$$

$$\frac{\left(\text{Sin}[c+d\,x]^j\right)^{m/2}\sqrt{b+a\,\text{Sin}[c+d\,x]}}{\sqrt{a+b\,\text{Csc}[c+d\,x]}}\int\frac{(b+a\,\text{Sin}[c+d\,x])^{n/2}}{\text{Sin}[c+d\,x]^{(n+1)/2}}\,dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_*(a_+b_.*sin[c_.+d_.*x_]^(-1))^n_,x_Symbol] :=
  Dist[(Sin[c+d*x]^j)^m*Sqrt[b+a*Sin[c+d*x]]/Sqrt[a+b*Csc[c+d*x]],
    Int[(b+a*sin[c+d*x])^n/sin[c+d*x]^(n+1/2),x]] /;
FreeQ[{a,b,c,d},x] && OneQ[j^2] && ZeroQ[a^2-b^2] && RationalQ[m,n] && j*m==-1/2 && n^2==1/4
```

$$\int \mathrm{Csc}[c+d\,x]^{m/2}\,(a+b\,\mathrm{Sin}[c+d\,x])^{n/2}\,dx$$

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{1}{\sqrt{1/z}\,\sqrt{a+b\,z}} = \frac{\sqrt{1/z}\,\sqrt{a+b\,z}}{b} - \frac{a\,\sqrt{1/z}}{b\,\sqrt{a+b\,z}}$
- **Rule: If** $a^2-b^2=0$, **then**

$$\int \frac{1}{\sqrt{\mathrm{Csc}[c+d\,x]}\,\sqrt{a+b\,\mathrm{Sin}[c+d\,x]}}\,dx \longrightarrow$$

$$\frac{1}{b}\int\sqrt{\mathrm{Csc}[c+d\,x]}\,\sqrt{a+b\,\mathrm{Sin}[c+d\,x]}\,dx-\frac{a}{b}\int\frac{\sqrt{\mathrm{Csc}[c+d\,x]}}{\sqrt{a+b\,\mathrm{Sin}[c+d\,x]}}\,dx$$

- **Program code:**

```
Int[1/(Sqrt[sin[c_.+d_.*x_]^(-1)]*Sqrt[a_+b_.*sin[c_.+d_.*x_]]),x_Symbol] :=
  Dist[1/b,Int[Sqrt[sin[c+d*x]^(-1)]*Sqrt[a+b*sin[c+d*x]],x]] -
  Dist[a/b,Int[Sqrt[sin[c+d*x]^(-1)]/Sqrt[a+b*sin[c+d*x]],x]] /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2-b^2]
```

- **Derivation: Piecewise constant extraction**
- **Basis:** $\partial_z\left(\sqrt{f[z]}\,\sqrt{f[z]^{-1}}\right)=0$
- **Rule: If** $a^2-b^2=0\,\bigwedge\,n^2=1$

$$\int\sqrt{\mathrm{Csc}[c+d\,x]}\,(a+b\,\mathrm{Sin}[c+d\,x])^{n/2}\,dx \longrightarrow$$

$$\sqrt{\mathrm{Csc}[c+d\,x]}\,\sqrt{\mathrm{Sin}[c+d\,x]}\int\frac{(a+b\,\mathrm{Sin}[c+d\,x])^{n/2}}{\sqrt{\mathrm{Sin}[c+d\,x]}}\,dx$$

- **Program code:**

```
Int[Sqrt[sin[c_.+d_.*x_]^(-1)]*(a_+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  Dist[Sqrt[Csc[c+d*x]]*Sqrt[Sin[c+d*x]],Int[(a+b*sin[c+d*x])^n/Sqrt[sin[c+d*x]],x]] /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2-b^2] && RationalQ[n] && n^2==1/4
```

Rules 9 – 10: $\int \left(\text{Sin}[c+d\,x]^j\right)^m \sqrt{a+b\,\text{Sin}[c+d\,x]^k}\,dx$

- **Derivation: Rule 9b with** $2\,j\,k\,m+k+2=0$

- **Rule 9a: If** $j^2=k^2=1\ \wedge\ a^2-b^2=0\ \wedge\ 2\,j\,k\,m+k+2=0$**, then**

$$\int \left(\text{Sin}[c+d\,x]^j\right)^m \sqrt{a+b\,\text{Sin}[c+d\,x]^k}\,dx\ \longrightarrow\ -\frac{2\,a\,\text{Cos}[c+d\,x]\,\left(\text{Sin}[c+d\,x]^j\right)^{m+j\,k}}{d\,\sqrt{a+b\,\text{Sin}[c+d\,x]^k}}$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_*Sqrt[a_+b_.*sin[c_.+d_.*x_]^k_.],x_Symbol] :=
  -2*a*Cos[c+d*x]*(Sin[c+d*x]^j)^(m+j*k)/(d*Sqrt[a+b*Sin[c+d*x]^k]) /;
FreeQ[{a,b,c,d,m},x] && OneQ[j^2,k^2] && ZeroQ[a^2-b^2] && ZeroQ[2*j*k*m+k+2]
```

- **Derivation: Rule 4b with** $n=\frac{1}{2}$

- **Rule 9b: If** $j^2=k^2=1\ \wedge\ a^2-b^2=0\ \wedge\ j\,k\,m\le-1\ \wedge\ 2\,j\,k\,m+k+2\ne0$**, then**

$$\int \left(\text{Sin}[c+d\,x]^j\right)^m \sqrt{a+b\,\text{Sin}[c+d\,x]^k}\,dx\ \longrightarrow$$

$$\frac{2\,a\,\text{Cos}[c+d\,x]\,\left(\text{Sin}[c+d\,x]^j\right)^{m+j\,k}}{d\,(2\,j\,k\,m+k+1)\,\sqrt{a+b\,\text{Sin}[c+d\,x]^k}}+\frac{b\,(2\,j\,k\,m+k+2)}{a\,(2\,j\,k\,m+k+1)}\int \left(\text{Sin}[c+d\,x]^j\right)^{m+j\,k} \sqrt{a+b\,\text{Sin}[c+d\,x]^k}\,dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*Sqrt[a_+b_.*sin[c_.+d_.*x_]^k_.],x_Symbol] :=
  2*a*Cos[c+d*x]*(Sin[c+d*x]^j)^(m+j*k)/(d*(2*j*k*m+k+1)*Sqrt[a+b*Sin[c+d*x]^k]) +
  Dist[b*(2*j*k*m+k+2)/(a*(2*j*k*m+k+1)),Int[(sin[c+d*x]^j)^(m+j*k)*Sqrt[a+b*sin[c+d*x]^k],x]] /;
FreeQ[{a,b,c,d},x] && OneQ[j^2,k^2] && ZeroQ[a^2-b^2] && RationalQ[m] &&
  j*k*m≤-1 && NonzeroQ[2*j*k*m+k+2]
```

- **Derivation: Rule 3b with** $n = \frac{1}{2}$

- **Rule 10: If** $j^2 = k^2 = 1 \land a^2 - b^2 = 0 \land 2 j m + 1 \neq 0 \land j k m > 0 \land j k m \neq 1 \land j k m \neq 2$**, then**

$$\int \left(\text{Sin}[c + d x]^j\right)^m \sqrt{a + b \text{Sin}[c + d x]^k} \, dx \longrightarrow$$

$$-\frac{2 b \text{Cos}[c + d x] \left(\text{Sin}[c + d x]^j\right)^m}{d k (2 j m + 1) \sqrt{a + b \text{Sin}[c + d x]^k}} + \frac{a (2 j k m + k - 1)}{b k (2 j m + 1)} \int \left(\text{Sin}[c + d x]^j\right)^{m - j k} \sqrt{a + b \text{Sin}[c + d x]^k} \, dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*Sqrt[a_+b_.*sin[c_.+d_.*x_]^k_.],x_Symbol] :=
  -2*b*Cos[c+d*x]*(Sin[c+d*x]^j)^m/(d*k*(2*j*m+1)*Sqrt[a+b*Sin[c+d*x]^k]) +
  Dist[a*(2*j*k*m+k-1)/(b*k*(2*j*m+1)),Int[(sin[c+d*x]^j)^(m-j*k)*Sqrt[a+b*sin[c+d*x]^k],x]] /;
FreeQ[{a,b,c,d},x] && OneQ[j^2,k^2] && ZeroQ[a^2-b^2] && RationalQ[m] &&
  NonzeroQ[2*j*m+1] && j*k*m>0 && j*k*m≠1 && j*k*m≠2
```

Rules 13 – 14: $\int \frac{\left(\operatorname{Sin}[c+d\,x]^j\right)^m}{\left(a+b\operatorname{Sin}[c+d\,x]^k\right)^{j\,k\,m+\frac{k+3}{2}}}\,dx$

- **Derivation: Rule 5 with** $j\,k\,m+n+\frac{k+3}{2}=0$
- **Rule 13: If** $j^2=k^2=1\wedge a^2-b^2=0\bigwedge j\,k\,m+n+\frac{k+3}{2}=0\wedge n>-1\wedge n\neq 1\wedge n\neq 2$**, then**

$$\int\left(\operatorname{Sin}[c+d\,x]^j\right)^m\left(a+b\operatorname{Sin}[c+d\,x]^k\right)^n dx\longrightarrow$$

$$-\frac{\operatorname{Cos}[c+d\,x]\left(\operatorname{Sin}[c+d\,x]^j\right)^{m+j\,k}\left(a+b\operatorname{Sin}[c+d\,x]^k\right)^n}{d\,(n+1)}+\frac{a\,n}{b\,(n+1)}\int\left(\operatorname{Sin}[c+d\,x]^j\right)^{m+j\,k}\left(a+b\operatorname{Sin}[c+d\,x]^k\right)^n dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  -Cos[c+d*x]*(Sin[c+d*x]^j)^(m+j*k)*(a+b*Sin[c+d*x]^k)^n/(d*(n+1)) +
  Dist[a*n/(b*(n+1)),Int[(sin[c+d*x]^j)^(m+j*k)*(a+b*sin[c+d*x]^k)^n,x]] /;
FreeQ[{a,b,c,d},x] && OneQ[j^2,k^2] && ZeroQ[a^2-b^2] && RationalQ[m,n] &&
  ZeroQ[j*k*m+n+(k+3)/2] && n>-1 && n≠1 && n≠2
```

- **Derivation: Rule 6 with** $j\,k\,m+n+\frac{k+3}{2}=0$
- **Rule 14: If** $j^2=k^2=1\wedge a^2-b^2=0\bigwedge j\,k\,m+n+\frac{k+3}{2}=0\wedge j\,k\,m\neq 1\wedge j\,k\,m\neq 2\wedge n<-1$**, then**

$$\int\left(\operatorname{Sin}[c+d\,x]^j\right)^m\left(a+b\operatorname{Sin}[c+d\,x]^k\right)^n dx\longrightarrow$$

$$-\frac{\operatorname{Cos}[c+d\,x]\left(\operatorname{Sin}[c+d\,x]^j\right)^{m+j\,k}\left(a+b\operatorname{Sin}[c+d\,x]^k\right)^n}{d\,(2\,n+1)}+\frac{n}{a\,(2\,n+1)}\int\left(\operatorname{Sin}[c+d\,x]^j\right)^m\left(a+b\operatorname{Sin}[c+d\,x]^k\right)^{n+1} dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  -Cos[c+d*x]*(Sin[c+d*x]^j)^(m+j*k)*(a+b*Sin[c+d*x]^k)^n/(d*(2*n+1)) +
  Dist[n/(a*(2*n+1)),Int[(sin[c+d*x]^j)^m*(a+b*sin[c+d*x]^k)^(n+1),x]] /;
FreeQ[{a,b,c,d},x] && OneQ[j^2,k^2] && ZeroQ[a^2-b^2] && RationalQ[m,n] &&
  ZeroQ[j*k*m+n+(k+3)/2] && j*k*m≠1 && j*k*m≠2 && n<-1
```

Rules 11 – 12: $\int \frac{\left(\operatorname{Sin}[c+d\,x]^j\right)^m}{\left(a+b\operatorname{Sin}[c+d\,x]^k\right)^{j\,k\,m+\frac{k+1}{2}}}\,dx$

- **Derivation: Rule 4b with** $j\,k\,m+n+\frac{k+1}{2}=0$
- **Rule 11: If** $j^2=k^2=1\wedge a^2-b^2=0\bigwedge j\,k\,m+n+\frac{k+1}{2}=0\bigwedge n>0\bigwedge n\neq\frac{1}{2}\bigwedge n\neq 1\bigwedge n\neq 2$**, then**

$$\int\left(\operatorname{Sin}[c+d\,x]^j\right)^m\left(a+b\operatorname{Sin}[c+d\,x]^k\right)^n dx\longrightarrow$$

$$-\frac{a\operatorname{Cos}[c+d\,x]\left(\operatorname{Sin}[c+d\,x]^j\right)^{m+j\,k}\left(a+b\operatorname{Sin}[c+d\,x]^k\right)^{n-1}}{d\,n}+$$

$$\frac{b\,(2\,n-1)}{n}\int\left(\operatorname{Sin}[c+d\,x]^j\right)^{m+j\,k}\left(a+b\operatorname{Sin}[c+d\,x]^k\right)^{n-1}dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  -a*Cos[c+d*x]*(Sin[c+d*x]^j)^(m+j*k)*(a+b*Sin[c+d*x]^k)^(n-1)/(d*n) +
  Dist[b*(2*n-1)/n,Int[(sin[c+d*x]^j)^(m+j*k)*(a+b*sin[c+d*x]^k)^(n-1),x]] /;
FreeQ[{a,b,c,d},x] && OneQ[j^2,k^2] && ZeroQ[a^2-b^2] && RationalQ[m,n] &&
  ZeroQ[j*k*m+n+(k+1)/2] && n>0 && n≠1/2 && n≠1 && n≠2
```

- **Derivation: Rule 1b with** $j\,k\,m+n+\frac{k+1}{2}=0$
- **Rule 12: If** $j^2=k^2=1\wedge a^2-b^2=0\bigwedge j\,k\,m+n+\frac{k+1}{2}=0\wedge j\,k\,m\neq 1\wedge j\,k\,m\neq 2\wedge n<-1$**, then**

$$\int\left(\operatorname{Sin}[c+d\,x]^j\right)^m\left(a+b\operatorname{Sin}[c+d\,x]^k\right)^n dx\longrightarrow$$

$$\frac{a\operatorname{Cos}[c+d\,x]\left(\operatorname{Sin}[c+d\,x]^j\right)^m\left(a+b\operatorname{Sin}[c+d\,x]^k\right)^n}{b\,d\,(2\,n+1)}+$$

$$\frac{n+1}{b\,(2\,n+1)}\int\left(\operatorname{Sin}[c+d\,x]^j\right)^{m-j\,k}\left(a+b\operatorname{Sin}[c+d\,x]^k\right)^{n+1}dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  a*Cos[c+d*x]*(Sin[c+d*x]^j)^m*(a+b*Sin[c+d*x]^k)^n/(b*d*(2*n+1)) +
  Dist[(n+1)/(b*(2*n+1)),Int[(sin[c+d*x]^j)^(m-j*k)*(a+b*sin[c+d*x]^k)^(n+1),x]] /;
FreeQ[{a,b,c,d},x] && OneQ[j^2,k^2] && ZeroQ[a^2-b^2] && RationalQ[m,n] &&
  ZeroQ[j*k*m+n+(k+1)/2] && j*k*m≠1 && j*k*m≠2 && n<-1
```

Rules 7 – 8: $\int \text{Sin}[c+d\,x]^{\frac{k-1}{2}}\left(a+b\,\text{Sin}[c+d\,x]^k\right)^n dx$

- **Reference: G&R 2.555.?**
- **Derivation: Rule 1b with** $j\,m = \frac{k-1}{2}$
- **Rule 7: If** $k^2 = 1 \land a^2 - b^2 = 0 \land n < -1$, **then**

$$\int \text{Sin}[c+d\,x]^{\frac{k-1}{2}}\left(a+b\,\text{Sin}[c+d\,x]^k\right)^n dx \longrightarrow$$

$$\frac{b\,\text{Cos}[c+d\,x]\,\text{Sin}[c+d\,x]^{\frac{k-1}{2}}\left(a+b\,\text{Sin}[c+d\,x]^k\right)^n}{a\,d\,(2\,n+1)} + \frac{n+1}{a\,(2\,n+1)}\int \text{Sin}[c+d\,x]^{\frac{k-1}{2}}\left(a+b\,\text{Sin}[c+d\,x]^k\right)^{n+1} dx$$

- **Program code:**

```
Int[(a_+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  b*Cos[c+d*x]*(a+b*Sin[c+d*x])^n/(a*d*(2*n+1)) +
  Dist[(n+1)/(a*(2*n+1)),Int[(a+b*sin[c+d*x])^(n+1),x]] /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2-b^2] && RationalQ[n] && n<-1
```

```
Int[sin[c_.+d_.*x_]^(-1)*(a_+b_.*sin[c_.+d_.*x_]^(-1))^n_,x_Symbol] :=
  b*Cot[c+d*x]*(a+b*Csc[c+d*x])^n/(a*d*(2*n+1)) +
   Dist[(n+1)/(a*(2*n+1)),Int[sin[c+d*x]^(-1)*(a+b/sin[c+d*x])^(n+1),x]] /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2-b^2] && RationalQ[n] && n<-1
```

- **Reference: G&R 2.555.? inverted**
- **Derivation: Rule 3b with** $j\,m = \frac{k-1}{2}$
- **Rule 8: If** $k^2 = 1 \land a^2 - b^2 = 0 \land n > 0 \land n \neq \frac{1}{2} \land n \neq 1 \land n \neq 2$, **then**

$$\int \text{Sin}[c+d\,x]^{\frac{k-1}{2}}\left(a+b\,\text{Sin}[c+d\,x]^k\right)^n dx \longrightarrow$$

$$-\frac{b\,\text{Cos}[c+d\,x]\,\text{Sin}[c+d\,x]^{\frac{k-1}{2}}\left(a+b\,\text{Sin}[c+d\,x]^k\right)^{n-1}}{d\,n} + \frac{a\,(2\,n-1)}{n}\int \text{Sin}[c+d\,x]^{\frac{k-1}{2}}\left(a+b\,\text{Sin}[c+d\,x]^k\right)^{n-1} dx$$

- **Program code:**

```
Int[(a_+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  -b*Cos[c+d*x]*(a+b*Sin[c+d*x])^(n-1)/(d*n) +
  Dist[a*(2*n-1)/n,Int[(a+b*sin[c+d*x])^(n-1),x]] /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2-b^2] && RationalQ[n] && n>0 && n≠1 && n≠2
```

```
Int[sin[c_.+d_.*x_]^(-1)*(a_+b_.*sin[c_.+d_.*x_]^(-1))^n_.,x_Symbol] :=
  -b*Cot[c+d*x]*(a+b*Csc[c+d*x])^(n-1)/(d*n) +
  Dist[a*(2*n-1)/n,Int[sin[c+d*x]^(-1)*(a+b*sin[c+d*x]^(-1))^(n-1),x]] /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2-b^2] && RationalQ[n] && n>0 && n≠1/2 && n≠1 && n≠2
```

Rules 1 – 6: $\int \left(\mathrm{Sin}[c+d\,x]^j\right)^m \left(a+b\,\mathrm{Sin}[c+d\,x]^k\right)^n dx$

- **Derivation: Recurrence 7 with $A = 0$, $B = 1$ and $m = m - 1$**
- **Rule 1a: If $j^2 = k^2 = 1 \wedge a^2 - b^2 = 0 \wedge j\,k\,m > 1 \wedge n \le -1$, then**

$$\int \left(\mathrm{Sin}[c+d\,x]^j\right)^m \left(a+b\,\mathrm{Sin}[c+d\,x]^k\right)^n dx \longrightarrow$$

$$-\frac{\mathrm{Cos}[c+d\,x] \left(\mathrm{Sin}[c+d\,x]^j\right)^{m-j\,k} \left(a+b\,\mathrm{Sin}[c+d\,x]^k\right)^n}{d\,(2\,n+1)} + \frac{1}{a^2\,(2\,n+1)} \cdot$$

$$\int \left(\mathrm{Sin}[c+d\,x]^j\right)^{m-2\,j\,k} \left(a\left(j\,k\,m+\frac{k-3}{2}\right) - b\left(j\,k\,m-n+\frac{k-3}{2}\right)\mathrm{Sin}[c+d\,x]^k\right) \left(a+b\,\mathrm{Sin}[c+d\,x]^k\right)^{n+1} dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  -Cos[c+d*x]*(Sin[c+d*x]^j)^(m-j*k)*(a+b*Sin[c+d*x]^k)^n/(d*(2*n+1)) +
  Dist[1/(a^2*(2*n+1)),
    Int[(sin[c+d*x]^j)^(m-2*j*k)*
      (a*(j*k*m+(k-3)/2)-b*(j*k*m-n+(k-3)/2)*sin[c+d*x]^k)*(a+b*sin[c+d*x]^k)^(n+1),x]] /;
FreeQ[{a,b,c,d},x] && OneQ[j^2,k^2] && ZeroQ[a^2-b^2] && RationalQ[m,n] &&
  j*k*m>1 && j*k*m≠2 && n≤-1
```

- **Derivation: Recurrence 7 with $A = 1$ and $B = 0$**
- **Derivation: Recurrence 12 with $A = 0$, $B = 1$ and $m = m - 1$**
- **Rule 1b: If $j^2 = k^2 = 1 \wedge a^2 - b^2 = 0 \wedge 0 < j\,k\,m < 1 \wedge n \le -1$, then**

$$\int \left(\mathrm{Sin}[c+d\,x]^j\right)^m \left(a+b\,\mathrm{Sin}[c+d\,x]^k\right)^n dx \longrightarrow$$

$$\frac{a\,\mathrm{Cos}[c+d\,x] \left(\mathrm{Sin}[c+d\,x]^j\right)^{m} \left(a+b\,\mathrm{Sin}[c+d\,x]^k\right)^n}{b\,d\,(2\,n+1)} + \frac{1}{b^2\,(2\,n+1)} \cdot$$

$$\int \left(\mathrm{Sin}[c+d\,x]^j\right)^{m-j\,k} \left(-b\left(j\,k\,m+\frac{k-1}{2}\right) + a\left(j\,k\,m+n+\frac{k+1}{2}\right)\mathrm{Sin}[c+d\,x]^k\right) \left(a+b\,\mathrm{Sin}[c+d\,x]^k\right)^{n+1} dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  a*Cos[c+d*x]*(Sin[c+d*x]^j)^m*(a+b*Sin[c+d*x]^k)^n/(b*d*(2*n+1)) +
  Dist[1/(b^2*(2*n+1)),
    Int[(sin[c+d*x]^j)^(m-j*k)*
      (-b*(j*k*m+(k-1)/2)+a*(j*k*m+n+(k+1)/2)*sin[c+d*x]^k)*(a+b*sin[c+d*x]^k)^(n+1),x]] /;
FreeQ[{a,b,c,d},x] && OneQ[j^2,k^2] && ZeroQ[a^2-b^2] && RationalQ[m,n] &&
  0<j*k*m<1 && n≤-1
```

- **Derivation: Recurrence 8 with $A = 0$, $B = 1$ and $m = m - 1$**

- **Rule 2: If $j^2 = k^2 = 1 \land a^2 - b^2 = 0 \bigwedge j k m + n + \frac{k-1}{2} \neq 0 \bigwedge j k m > 1 \bigwedge -1 < n < 0 \bigwedge j k m - 1 \neq n$, then**

$$\int \left(\text{Sin}[c + d x]^j\right)^m \left(a + b \,\text{Sin}[c + d x]^k\right)^n dx \longrightarrow$$

$$-\frac{\text{Cos}[c + d x] \left(\text{Sin}[c + d x]^j\right)^{m-j k} \left(a + b \,\text{Sin}[c + d x]^k\right)^n}{d \left(j k m + n + \frac{k-1}{2}\right)} + \frac{1}{b \left(j k m + n + \frac{k-1}{2}\right)} \cdot$$

$$\int \left(\text{Sin}[c + d x]^j\right)^{m-2 j k} \left(b \left(j k m + \frac{k-3}{2}\right) + a n \,\text{Sin}[c + d x]^k\right) \left(a + b \,\text{Sin}[c + d x]^k\right)^n dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  -Cos[c+d*x]*(Sin[c+d*x]^j)^(m-j*k)*(a+b*Sin[c+d*x]^k)^n/(d*(j*k*m+n+(k-1)/2)) +
  Dist[1/(b*(j*k*m+n+(k-1)/2)),
    Int[(sin[c+d*x]^j)^(m-2*j*k)*(b*(j*k*m+(k-3)/2)+a*n*sin[c+d*x]^k)*(a+b*sin[c+d*x]^k)^n,x]] /;
FreeQ[{a,b,c,d},x] && OneQ[j^2,k^2] && ZeroQ[a^2-b^2] && RationalQ[m,n] &&
  NonzeroQ[j*k*m+n+(k-1)/2] && j*k*m>1 && j*k*m≠2 && -1<n<0 && j*k*m-1≠n
```

- **Derivation: Recurrence 9 with $A = a$, $B = b$ and $n = n - 1$**

- **Rule 3a: If $j^2 = k^2 = 1 \land a^2 - b^2 = 0 \bigwedge j k m + n + \frac{k-1}{2} \neq 0 \bigwedge j k m \geq -1 \bigwedge j k m \neq 1 \bigwedge j k m \neq 2 \bigwedge n > 1 \bigwedge n \neq 2$, then**

$$\int \left(\text{Sin}[c + d x]^j\right)^m \left(a + b \,\text{Sin}[c + d x]^k\right)^n dx \longrightarrow$$

$$-\frac{b^2 \,\text{Cos}[c + d x] \left(\text{Sin}[c + d x]^j\right)^{m+j k} \left(a + b \,\text{Sin}[c + d x]^k\right)^{n-2}}{d \left(j k m + n + \frac{k-1}{2}\right)} + \frac{a}{j k m + n + \frac{k-1}{2}} \cdot$$

$$\int \left(\text{Sin}[c + d x]^j\right)^m \left(a (2 j k m + n + k) + b (2 j k m + 3 n + k - 3) \,\text{Sin}[c + d x]^k\right) \left(a + b \,\text{Sin}[c + d x]^k\right)^{n-2} dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  -b^2*Cos[c+d*x]*(Sin[c+d*x]^j)^(m+j*k)*(a+b*Sin[c+d*x]^k)^(n-2)/(d*(j*k*m+n+(k-1)/2)) +
  Dist[a/(j*k*m+n+(k-1)/2),
    Int[(sin[c+d*x]^j)^m*
      (a*(2*j*k*m+n+k)+b*(2*j*k*m+3*n+k-3)*sin[c+d*x]^k)*(a+b*sin[c+d*x]^k)^(n-2),x]] /;
FreeQ[{a,b,c,d},x] && OneQ[j^2,k^2] && ZeroQ[a^2-b^2] && RationalQ[m,n] &&
  NonzeroQ[j*k*m+n+(k-1)/2] && j*k*m≥-1 && j*k*m≠1 && j*k*m≠2 && n>1 && n≠2
```

- **Derivation: Recurrence 8 with** $A = a$, $B = b$ **and** $n = n - 1$
- **Derivation: Recurrence 9 with** $A = 0$, $B = 1$ **and** $m = m - 1$
- **Note: In the case** $n = \frac{1}{2}$**, this rule simplifies to rule 10.**
- **Rule 3b: If** $j^2 = k^2 = 1 \wedge a^2 - b^2 = 0 \bigwedge j\,k\,m > 0 \bigwedge j\,k\,m \neq 1 \bigwedge j\,k\,m \neq 2 \bigwedge 0 < n < 1 \bigwedge n \neq \frac{1}{2}$**, then**

$$\int \left(\text{Sin}[c+d\,x]^j\right)^m \left(a+b\,\text{Sin}[c+d\,x]^k\right)^n dx \longrightarrow$$

$$-\frac{b\,\text{Cos}[c+d\,x] \left(\text{Sin}[c+d\,x]^j\right)^m \left(a+b\,\text{Sin}[c+d\,x]^k\right)^{n-1}}{d\left(j\,k\,m+n+\frac{k-1}{2}\right)}+\frac{1}{j\,k\,m+n+\frac{k-1}{2}}\cdot$$
$$\int \left(\text{Sin}[c+d\,x]^j\right)^{m-j\,k} \left(b\left(j\,k\,m+\frac{k-1}{2}\right)+a\left(j\,k\,m+2\,n+\frac{k-3}{2}\right)\text{Sin}[c+d\,x]^k\right)\left(a+b\,\text{Sin}[c+d\,x]^k\right)^{n-1} dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  -b*Cos[c+d*x]*(Sin[c+d*x]^j)^m*(a+b*Sin[c+d*x]^k)^(n-1)/(d*(j*k*m+n+(k-1)/2)) +
  Dist[1/(j*k*m+n+(k-1)/2),
    Int[(sin[c+d*x]^j)^(m-j*k)*
      (b*(j*k*m+(k-1)/2)+a*(j*k*m+2*n+(k-3)/2)*sin[c+d*x]^k)*(a+b*sin[c+d*x]^k)^(n-1),x]] /;
FreeQ[{a,b,c,d},x] && OneQ[j^2,k^2] && ZeroQ[a^2-b^2] && RationalQ[m,n] &&
  j*k*m>0 && j*k*m≠1 && j*k*m≠2 && 0<n<1 && n≠1/2
```

- **Derivation: Recurrence 10 with** $A = a$, $B = b$ **and** $n = n - 1$
- **Rule 4a: If** $j^2 = k^2 = 1 \wedge a^2 - b^2 = 0 \wedge j\,k\,m < -1 \wedge n > 1 \wedge n \neq 2$**, then**

$$\int \left(\text{Sin}[c+d\,x]^j\right)^m \left(a+b\,\text{Sin}[c+d\,x]^k\right)^n dx \longrightarrow$$

$$\frac{a^2\,\text{Cos}[c+d\,x] \left(\text{Sin}[c+d\,x]^j\right)^{m+j\,k} \left(a+b\,\text{Sin}[c+d\,x]^k\right)^{n-2}}{d\left(j\,k\,m+\frac{k+1}{2}\right)}+\frac{a}{j\,k\,m+\frac{k+1}{2}}\cdot$$
$$\int \left(\text{Sin}[c+d\,x]^j\right)^{m+j\,k} \left(b\,(2\,j\,k\,m-n+k+3)+a\,(2\,j\,k\,m+n+k)\,\text{Sin}[c+d\,x]^k\right)\left(a+b\,\text{Sin}[c+d\,x]^k\right)^{n-2} dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  a^2*Cos[c+d*x]*(Sin[c+d*x]^j)^(m+j*k)*(a+b*Sin[c+d*x]^k)^(n-2)/(d*(j*k*m+(k+1)/2)) +
  Dist[a/(j*k*m+(k+1)/2),
    Int[(sin[c+d*x]^j)^(m+j*k)*
      (b*(2*j*k*m-n+k+3)+a*(2*j*k*m+n+k)*sin[c+d*x]^k)*(a+b*sin[c+d*x]^k)^(n-2),x]] /;
FreeQ[{a,b,c,d},x] && OneQ[j^2,k^2] && ZeroQ[a^2-b^2] && RationalQ[m,n] &&
  j*k*m<-1 && n>1 && n≠2
```

- **Derivation: Recurrence 10 with** $A = 1$ **and** $B = 0$
- **Derivation: Recurrence 11 with** $A = a$, $B = b$ **and** $n = n - 1$
- **Note: In the case** $n = \frac{1}{2}$**, this rule simplifies to rule 9b.**
- **Rule 4b: If** $j^2 = k^2 = 1 \wedge a^2 - b^2 = 0 \bigwedge j\,k\,m < -1 \bigwedge 0 < n < 1 \bigwedge n \neq \frac{1}{2}$**, then**

$$\int \left(\text{Sin}[c + d\,x]^j\right)^m \left(a + b\,\text{Sin}[c + d\,x]^k\right)^n dx \longrightarrow$$

$$\frac{a\,\text{Cos}[c + d\,x] \left(\text{Sin}[c + d\,x]^j\right)^{m+j\,k} \left(a + b\,\text{Sin}[c + d\,x]^k\right)^{n-1}}{d \left(j\,k\,m + \frac{k+1}{2}\right)} + \frac{1}{j\,k\,m + \frac{k+1}{2}} \cdot$$

$$\int \left(\text{Sin}[c + d\,x]^j\right)^{m+j\,k} \left(b \left(j\,k\,m - n + \frac{k+3}{2}\right) + a \left(j\,k\,m + n + \frac{k+1}{2}\right) \text{Sin}[c + d\,x]^k\right) \left(a + b\,\text{Sin}[c + d\,x]^k\right)^{n-1} dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  a*Cos[c+d*x]*(Sin[c+d*x]^j)^(m+j*k)*(a+b*Sin[c+d*x]^k)^(n-1)/(d*(j*k*m+(k+1)/2)) +
  Dist[1/(j*k*m+(k+1)/2),
    Int[(sin[c+d*x]^j)^(m+j*k)*
      (b*(j*k*m-n+(k+3)/2)+a*(j*k*m+n+(k+1)/2)*sin[c+d*x]^k)*(a+b*sin[c+d*x]^k)^(n-1),x]] /;
FreeQ[{a,b,c,d},x] && OneQ[j^2,k^2] && ZeroQ[a^2-b^2] && RationalQ[m,n] &&
  j*k*m<-1 && 0<n<1 && n≠1/2
```

- **Derivation: Recurrence 11 with** $A = 1$ **and** $B = 0$
- **Rule 5: If** $j^2 = k^2 = 1 \wedge a^2 - b^2 = 0 \bigwedge j\,k\,m + \frac{k+1}{2} \neq 0 \bigwedge j\,k\,m \leq -1 \bigwedge -1 < n < 0$**, then**

$$\int \left(\text{Sin}[c + d\,x]^j\right)^m \left(a + b\,\text{Sin}[c + d\,x]^k\right)^n dx \longrightarrow$$

$$\frac{\text{Cos}[c + d\,x] \left(\text{Sin}[c + d\,x]^j\right)^{m+j\,k} \left(a + b\,\text{Sin}[c + d\,x]^k\right)^{n}}{d \left(j\,k\,m + \frac{k+1}{2}\right)} + \frac{1}{b \left(j\,k\,m + \frac{k+1}{2}\right)} \cdot$$

$$\int \left(\text{Sin}[c + d\,x]^j\right)^{m+j\,k} \left(-a\,n + b \left(j\,k\,m + n + \frac{k+3}{2}\right) \text{Sin}[c + d\,x]^k\right) \left(a + b\,\text{Sin}[c + d\,x]^k\right)^{n} dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  Cos[c+d*x]*(Sin[c+d*x]^j)^(m+j*k)*(a+b*Sin[c+d*x]^k)^n/(d*(j*k*m+(k+1)/2)) +
  Dist[1/(b*(j*k*m+(k+1)/2)),
    Int[(sin[c+d*x]^j)^(m+j*k)*(-a*n+b*(j*k*m+n+(k+3)/2)*sin[c+d*x]^k)*(a+b*sin[c+d*x]^k)^n,x]] /;
FreeQ[{a,b,c,d},x] && OneQ[j^2,k^2] && ZeroQ[a^2-b^2] && RationalQ[m,n] &&
  NonzeroQ[j*k*m+(k+1)/2] && j*k*m≤-1 && -1<n<0
```

- **Derivation: Recurrence 12 with $A = 1$ and $B = 0$**
- **Rule 6: If $j^2 = k^2 = 1 \wedge a^2 - b^2 = 0 \wedge j\,k\,m < 0 \wedge n \le -1$, then**

$$\int \left(\text{Sin}[c+d\,x]^j\right)^m \left(a+b\,\text{Sin}[c+d\,x]^k\right)^n dx \longrightarrow$$

$$-\frac{\text{Cos}[c+d\,x]\left(\text{Sin}[c+d\,x]^j\right)^{m+j\,k}\left(a+b\,\text{Sin}[c+d\,x]^k\right)^n}{d\,(2\,n+1)}+\frac{1}{a^2\,(2\,n+1)}\cdot$$

$$\int \left(\text{Sin}[c+d\,x]^j\right)^m \left(a\left(j\,k\,m+2\,n+\frac{k+3}{2}\right)-b\left(j\,k\,m+n+\frac{k+3}{2}\right)\text{Sin}[c+d\,x]^k\right)\left(a+b\,\text{Sin}[c+d\,x]^k\right)^{n+1} dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  -Cos[c+d*x]*(Sin[c+d*x]^j)^(m+j*k)*(a+b*Sin[c+d*x]^k)^n/(d*(2*n+1)) +
  Dist[1/(a^2*(2*n+1)),
    Int[(sin[c+d*x]^j)^m*
      (a*(j*k*m+2*n+(k+3)/2)-b*(j*k*m+n+(k+3)/2)*sin[c+d*x]^k)*(a+b*sin[c+d*x]^k)^(n+1),x]] /;
FreeQ[{a,b,c,d},x] && OneQ[j^2,k^2] && ZeroQ[a^2-b^2] && RationalQ[m,n] &&
  j*k*m<0 && n≤-1
```