$$\int (c\,\text{ProductLog}[a+b\,x])^p\,dx$$

- **Rule: If $p<-1$, then**

$$\int (c\,\text{ProductLog}[a+b\,x])^p\,dx \longrightarrow \frac{(a+b\,x)\,(c\,\text{ProductLog}[a+b\,x])^p}{b\,(p+1)}+\frac{p}{c\,(p+1)}\int\frac{(c\,\text{ProductLog}[a+b\,x])^{p+1}}{1+\text{ProductLog}[a+b\,x]}\,dx$$

- **Program code:**

```
Int[(c_.*ProductLog[a_.+b_.*x_])^p_,x_Symbol] :=
  (a+b*x)*(c*ProductLog[a+b*x])^p/(b*(p+1)) +
  Dist[p/(c*(p+1)),Int[(c*ProductLog[a+b*x])^(p+1)/(1+ProductLog[a+b*x]),x]] /;
FreeQ[{a,b,c},x] && RationalQ[p] && p<-1
```

- **Derivation: Integration by parts**
- **Rule: If $\neg\,(p<-1)$, then**

$$\int (c\,\text{ProductLog}[a+b\,x])^p\,dx \longrightarrow \frac{(a+b\,x)\,(c\,\text{ProductLog}[a+b\,x])^p}{b}-p\int\frac{(c\,\text{ProductLog}[a+b\,x])^{p}}{1+\text{ProductLog}[a+b\,x]}\,dx$$

- **Program code:**

```
Int[(c_.*ProductLog[a_.+b_.*x_])^p_.,x_Symbol] :=
  (a+b*x)*(c*ProductLog[a+b*x])^p/b -
  Dist[p,Int[(c*ProductLog[a+b*x])^p/(1+ProductLog[a+b*x]),x]] /;
FreeQ[{a,b,c},x] && Not[RationalQ[p] && p<-1]
```

$$\int \frac{1}{d + d\,\text{ProductLog}[a + b\,x]}\,dx$$

- **Rule:**

$$\int \frac{1}{d + d\,\text{ProductLog}[a + b\,x]}\,dx \longrightarrow \frac{a + b\,x}{b\,d\,\text{ProductLog}[a + b\,x]}$$

- **Program code:**

```
Int[1/(d_+d_.*ProductLog[a_.+b_.*x_]),x_Symbol] :=
  (a+b*x)/(b*d*ProductLog[a+b*x]) /;
FreeQ[{a,b,d},x]
```

$$\int \frac{(c\,\text{ProductLog}[a+b\,x])^p}{d+d\,\text{ProductLog}[a+b\,x]}\,dx$$

- **Derivation: Algebraic simplification**
- **Basis:** $\frac{z}{1+z} = 1 - \frac{1}{1+z}$
- **Rule:**

$$\int \frac{\text{ProductLog}[a+b\,x]}{d+d\,\text{ProductLog}[a+b\,x]}\,dx \longrightarrow d\,x - \int \frac{1}{d+d\,\text{ProductLog}[a+b\,x]}\,dx$$

- **Program code:**

```
Int[ProductLog[a_.+b_.*x_]/(d_+d_.*ProductLog[a_.+b_.*x_]),x_Symbol] :=
  d*x -
  Int[1/(d+d*ProductLog[a+b*x]),x] /;
FreeQ[{a,b,d},x]
```

- **Rule:**

$$\int \frac{1}{\text{ProductLog}[a+b\,x]\,(d+d\,\text{ProductLog}[a+b\,x])}\,dx \longrightarrow \frac{\text{ExpIntegralEi}[\text{ProductLog}[a+b\,x]]}{b\,d}$$

- **Program code:**

```
Int[1/(ProductLog[a_.+b_.*x_]*(d_+d_.*ProductLog[a_.+b_.*x_])),x_Symbol] :=
  ExpIntegralEi[ProductLog[a+b*x]]/(b*d) /;
FreeQ[{a,b,d},x]
```

- **Rule: If $c > 0$, then**

$$\int \frac{1}{\text{Sqrt}[c\,\text{ProductLog}[a+b\,x]]\,(d+d\,\text{ProductLog}[a+b\,x])}\,dx \longrightarrow \frac{\sqrt{\pi\,c}}{b\,c\,d}\,\text{Erfi}\left[\frac{\sqrt{c\,\text{ProductLog}[a+b\,x]}}{\sqrt{c}}\right]$$

- **Program code:**

```
Int[1/(Sqrt[c_.*ProductLog[a_.+b_.*x_]]*(d_+d_.*ProductLog[a_.+b_.*x_])),x_Symbol] :=
  Rt[Pi*c,2]*Erfi[Sqrt[c*ProductLog[a+b*x]]/Rt[c,2]]/(b*c*d) /;
FreeQ[{a,b,c,d},x] && PosQ[c]
```

- **Rule: If $c < 0$, then**

$$\int \frac{1}{\text{Sqrt}[c\,\text{ProductLog}[a+b\,x]]\,(d+d\,\text{ProductLog}[a+b\,x])}\,dx \longrightarrow \frac{\sqrt{-\pi\,c}}{b\,c\,d}\,\text{Erf}\Big[\frac{\sqrt{c\,\text{ProductLog}[a+b\,x]}}{\sqrt{-c}}\Big]$$

- **Program code:**

```
Int[1/(Sqrt[c_.*ProductLog[a_.+b_.*x_]]*(d_+d_.*ProductLog[a_.+b_.*x_])),x_Symbol] :=
  Rt[-Pi*c,2]*Erf[Sqrt[c*ProductLog[a+b*x]]/Rt[-c,2]]/(b*c*d) /;
FreeQ[{a,b,c,d},x] && NegQ[c]
```

- **Rule: If $p > 0$, then**

$$\int \frac{(c\,\text{ProductLog}[a+b\,x])^p}{d+d\,\text{ProductLog}[a+b\,x]}\,dx \longrightarrow$$
$$\frac{c\,(a+b\,x)\,(c\,\text{ProductLog}[a+b\,x])^{p-1}}{b\,d} - c\,p\int \frac{(c\,\text{ProductLog}[a+b\,x])^{p-1}}{d+d\,\text{ProductLog}[a+b\,x]}\,dx$$

- **Program code:**

```
Int[(c_.*ProductLog[a_.+b_.*x_])^p_/(d_+d_.*ProductLog[a_.+b_.*x_]),x_Symbol] :=
  c*(a+b*x)*(c*ProductLog[a+b*x])^(p-1)/(b*d) -
  Dist[c*p,Int[(c*ProductLog[a+b*x])^(p-1)/(d+d*ProductLog[a+b*x]),x]] /;
FreeQ[{a,b,c,d},x] && RationalQ[p] && p>0
```

- **Rule: If $p < -1$, then**

$$\int \frac{(c\,\text{ProductLog}[a+b\,x])^p}{d+d\,\text{ProductLog}[a+b\,x]}\,dx \longrightarrow$$
$$\frac{(a+b\,x)\,(c\,\text{ProductLog}[a+b\,x])^p}{b\,d\,(p+1)} - \frac{1}{c\,(p+1)}\int \frac{(c\,\text{ProductLog}[a+b\,x])^{p+1}}{d+d\,\text{ProductLog}[a+b\,x]}\,dx$$

- **Program code:**

```
Int[(c_.*ProductLog[a_.+b_.*x_])^p_/(d_+d_.*ProductLog[a_.+b_.*x_]),x_Symbol] :=
  (a+b*x)*(c*ProductLog[a+b*x])^p/(b*d*(p+1)) -
  Dist[1/(c*(p+1)),Int[(c*ProductLog[a+b*x])^(p+1)/(d+d*ProductLog[a+b*x]),x]] /;
FreeQ[{a,b,c,d},x] && RationalQ[p] && p<-1
```

- **Rule:**

$$\int \frac{(c\,\text{ProductLog}[a+b\,x])^p}{d+d\,\text{ProductLog}[a+b\,x]}\,dx \longrightarrow \frac{\text{Gamma}[p+1,\,-\text{ProductLog}[a+b\,x]]\,(c\,\text{ProductLog}[a+b\,x])^p}{b\,d\,(-\text{ProductLog}[a+b\,x])^p}$$

- **Program code:**

```
Int[(c_.*ProductLog[a_.+b_.*x_])^p_./(d_+d_.*ProductLog[a_.+b_.*x_]),x_Symbol] :=
  Gamma[p+1,-ProductLog[a+b*x]]*(c*ProductLog[a+b*x])^p/(b*d*(-ProductLog[a+b*x])^p) /;
FreeQ[{a,b,c,d,p},x]
```

$$\int x^m\,(c\,\text{ProductLog}[a+b\,x])^p\,dx$$

- **Derivation: Integration by substitution**
- **Rule: If** $m \in \mathbb{Z} \land m > 0$**, then**

$$\int x^m\,(c\,\text{ProductLog}[a+b\,x])^p\,dx \longrightarrow \frac{1}{b}\,\text{Subst}\left[\int\left(-\frac{a}{b}+\frac{x}{b}\right)^m\,(c\,\text{ProductLog}[x])^p\,dx,\ x,\ a+b\,x\right]$$

- **Program code:**

```
Int[x_^m_.*(c_.*ProductLog[a_+b_.*x_])^p_.,x_Symbol] :=
  Dist[1/b,Subst[Int[Dist[(c*ProductLog[x])^p,Expand[(-a/b+x/b)^m]],x],x,a+b*x]] /;
FreeQ[{a,b,c,p},x] && IntegerQ[m] && m>0
```

$$\int \frac{x^m}{d + d\,\text{ProductLog}[a + b\,x]}\,dx$$

- **Derivation: Integration by substitution**
- **Rule: If** $m \in \mathbb{Z} \land m > 0$, **then**

$$\int \frac{x^m}{d + d\,\text{ProductLog}[a + b\,x]}\,dx \longrightarrow \frac{1}{b}\,\text{Subst}\left[\int \frac{\left(-\frac{a}{b} + \frac{x}{b}\right)^m}{d + d\,\text{ProductLog}[x]}\,dx,\ x,\ a + b\,x\right]$$

- **Program code:**

```
Int[x_^m_./(d_+d_.*ProductLog[a_+b_.*x_]),x_Symbol] :=
  Dist[1/b,Subst[Int[Dist[1/(d+d*ProductLog[x]),Expand[(-a/b+x/b)^m]],x],x,a+b*x]] /;
FreeQ[{a,b,d},x] && IntegerQ[m] && m>0
```

$$\int \frac{x^m \, (c \, \text{ProductLog}[a + b \, x])^p}{d + d \, \text{ProductLog}[a + b \, x]} \, dx$$

- **Derivation: Integration by substitution**
- **Rule: If** $m \in \mathbb{Z} \wedge m > 0$**, then**

$$\int \frac{x^m \, (c \, \text{ProductLog}[a + b \, x])^p}{d + d \, \text{ProductLog}[a + b \, x]} \, dx \longrightarrow \frac{1}{b} \, \text{Subst}\left[\int \frac{\left(-\frac{a}{b} + \frac{x}{b}\right)^m \, (c \, \text{ProductLog}[x])^p}{d + d \, \text{ProductLog}[x]} \, dx, \, x, \, a + b \, x\right]$$

- **Program code:**

```
Int[x_^m_.*(c_.*ProductLog[a_+b_.*x_])^p_./(d_+d_.*ProductLog[a_+b_.*x_]),x_Symbol] :=
  Dist[1/b,Subst[Int[Dist[(c*ProductLog[x])^p/(d+d*ProductLog[x]),Expand[(-a/b+x/b)^m]],x],x,a+b*x]] /;
FreeQ[{a,b,c,d,p},x] && IntegerQ[m] && m>0
```

$$\int (c\,\mathrm{ProductLog}[a\,x^n])^p\,dx$$

- **Derivation: Integration by parts**
- **Rule: If** $n\,(p-1)+1=0\ \bigvee\ \left(p-\frac{1}{2}\in\mathbb{Z}\ \bigwedge\ n\left(p-\frac{1}{2}\right)+1=0\right)$**, then**

$$\int (c\,\mathrm{ProductLog}[a\,x^n])^p\,dx \longrightarrow x\,(c\,\mathrm{ProductLog}[a\,x^n])^p - n\,p\int \frac{(c\,\mathrm{ProductLog}[a\,x^n])^p}{1+\mathrm{ProductLog}[a\,x^n]}\,dx$$

- **Program code:**

```
Int[(c_.*ProductLog[a_.*x_^n_])^p_.,x_Symbol] :=
  x*(c*ProductLog[a*x^n])^p -
  Dist[n*p,Int[(c*ProductLog[a*x^n])^p/(1+ProductLog[a*x^n]),x]] /;
FreeQ[{a,c,n,p},x] && (ZeroQ[n*(p-1)+1] || IntegerQ[p-1/2] && ZeroQ[n*(p-1/2)+1])
```

- **Rule: If** $(p\in\mathbb{Z}\ \wedge\ n\,(p+1)+1=0)\ \bigvee\ \left(p-\frac{1}{2}\in\mathbb{Z}\ \bigwedge\ n\left(p+\frac{1}{2}\right)+1=0\right)$**, then**

$$\int (c\,\mathrm{ProductLog}[a\,x^n])^p\,dx \longrightarrow \frac{x\,(c\,\mathrm{ProductLog}[a\,x^n])^p}{n\,p+1} + \frac{n\,p}{c\,(n\,p+1)}\int \frac{(c\,\mathrm{ProductLog}[a\,x^n])^{p+1}}{1+\mathrm{ProductLog}[a\,x^n]}\,dx$$

- **Program code:**

```
Int[(c_.*ProductLog[a_.*x_^n_])^p_.,x_Symbol] :=
  x*(c*ProductLog[a*x^n])^p/(n*p+1) +
  Dist[n*p/(c*(n*p+1)),Int[(c*ProductLog[a*x^n])^(p+1)/(1+ProductLog[a*x^n]),x]] /;
FreeQ[{a,c,n},x] && (IntegerQ[p] && ZeroQ[n*(p+1)+1] || IntegerQ[p-1/2] && ZeroQ[n*(p+1/2)+1])
```

- **Derivation: Integration by substitution**
- **Basis:** $\int f[x]\,dx = -\mathrm{Subst}\left[\int \frac{f\left[\frac{1}{x}\right]}{x^2}\,dx,\ x,\ \frac{1}{x}\right]$
- **Rule: If** $n\in\mathbb{Z}\ \wedge\ n<0$**, then**

$$\int (c\,\mathrm{ProductLog}[a\,x^n])^p\,dx \longrightarrow -\mathrm{Subst}\left[\int \frac{(c\,\mathrm{ProductLog}[a\,x^{-n}])^p}{x^2}\,dx,\ x,\ \frac{1}{x}\right]$$

- **Program code:**

```
Int[(c_.*ProductLog[a_.*x_^n_])^p_.,x_Symbol] :=
  -Subst[Int[(c*ProductLog[a*x^(-n)])^p/x^2,x],x,1/x] /;
FreeQ[{a,c,p},x] && IntegerQ[n] && n<0
```

$$\int \frac{1}{d + d\,\text{ProductLog}[a\,x^n]}\,dx$$

- **Derivation: Integration by substitution**

- **Basis:** $\int f[x]\,dx = -\text{Subst}\left[\int \frac{f\left[\frac{1}{x}\right]}{x^2}\,dx,\ x,\ \frac{1}{x}\right]$

- **Rule: If** $n \in \mathbb{Z} \wedge n < 0$**, then**

$$\int \frac{1}{d + d\,\text{ProductLog}[a\,x^n]}\,dx \longrightarrow -\text{Subst}\left[\int \frac{1}{x^2\,(d + d\,\text{ProductLog}[a\,x^{-n}])}\,dx,\ x,\ \frac{1}{x}\right]$$

- **Program code:**

```
Int[1/(d_+d_.*ProductLog[a_.*x_^n_]),x_Symbol] :=
  -Subst[Int[1/(x^2*(d+d*ProductLog[a*x^(-n)])),x],x,1/x] /;
FreeQ[{a,d},x] && IntegerQ[n] && n<0
```

$$\int \frac{(c\,\text{ProductLog}[a\,x^n])^p}{d + d\,\text{ProductLog}[a\,x^n]}\,dx$$

- **Rule: If** $n\,(p-1)+1=0$**, then**

$$\int \frac{(c\,\text{ProductLog}[a\,x^n])^p}{d + d\,\text{ProductLog}[a\,x^n]}\,dx \longrightarrow \frac{c\,x\,(c\,\text{ProductLog}[a\,x^n])^{p-1}}{d}$$

- **Program code:**

```
Int[(c_.*ProductLog[a_.*x_^n_.])^p_./(d_+d_.*ProductLog[a_.*x_^n_.]),x_Symbol] :=
  c*x*(c*ProductLog[a*x^n])^(p-1)/d /;
FreeQ[{a,c,d,n,p},x] && ZeroQ[n*(p-1)+1]
```

- **Rule: If** $p \in \mathbb{Z} \bigwedge n = -\frac{1}{p}$**, then**

$$\int \frac{\text{ProductLog}[a\,x^n]^p}{d + d\,\text{ProductLog}[a\,x^n]}\,dx \longrightarrow \frac{a^p\,\text{ExpIntegralEi}[-p\,\text{ProductLog}[a\,x^n]]}{d\,n}$$

- **Program code:**

```
Int[ProductLog[a_.*x_^n_.]^p_./(d_+d_.*ProductLog[a_.*x_^n_.]),x_Symbol] :=
  a^p*ExpIntegralEi[-p*ProductLog[a*x^n]]/(d*n) /;
FreeQ[{a,d},x] && IntegerQ[1/n] && ZeroQ[p+1/n]
```

- **Rule: If** $\frac{1}{n} \in \mathbb{Z} \bigwedge p = \frac{1}{2} - \frac{1}{n} \bigwedge c\,n > 0$**, then**

$$\int \frac{(c\,\text{ProductLog}[a\,x^n])^p}{d + d\,\text{ProductLog}[a\,x^n]}\,dx \longrightarrow \frac{\sqrt{\pi\,c\,n}}{d\,n\,a^{1/n}\,c^{1/n}}\,\text{Erfi}\left[\frac{\sqrt{c\,\text{ProductLog}[a\,x^n]}}{\sqrt{c\,n}}\right]$$

- **Program code:**

```
Int[(c_.*ProductLog[a_.*x_^n_.])^p_/(d_+d_.*ProductLog[a_.*x_^n_.]),x_Symbol] :=
  Rt[Pi*c*n,2]/(d*n*a^(1/n)*c^(1/n))*Erfi[Sqrt[c*ProductLog[a*x^n]]/Rt[c*n,2]] /;
FreeQ[{a,c,d},x] && IntegerQ[1/n] && ZeroQ[p-1/2+1/n] && PosQ[c*n]
```

- **Rule: If** $\frac{1}{n} \in \mathbb{Z} \bigwedge p = \frac{1}{2} - \frac{1}{n} \bigwedge c\, n < 0$, **then**

$$\int \frac{(c \operatorname{ProductLog}[a x^n])^p}{d + d \operatorname{ProductLog}[a x^n]} \, dx \longrightarrow \frac{\sqrt{-\pi c n}}{d\, n\, a^{1/n} c^{1/n}} \operatorname{Erf}\left[\frac{\sqrt{c \operatorname{ProductLog}[a x^n]}}{\sqrt{-c n}}\right]$$

- **Program code:**

```
Int[(c_.*ProductLog[a_.*x_^n_.])^p_/(d_+d_.*ProductLog[a_.*x_^n_.]),x_Symbol] :=
  Rt[-Pi*c*n,2]/(d*n*a^(1/n)*c^(1/n))*Erf[Sqrt[c*ProductLog[a*x^n]]/Rt[-c*n,2]] /;
FreeQ[{a,c,d},x] && IntegerQ[1/n] && ZeroQ[p-1/2+1/n] && NegQ[c*n]
```

- **Rule: If** $n > 0 \bigwedge n\,(p-1)+1 > 0$, **then**

$$\int \frac{(c \operatorname{ProductLog}[a x^n])^p}{d + d \operatorname{ProductLog}[a x^n]} \, dx \longrightarrow \frac{c\, x\, (c \operatorname{ProductLog}[a x^n])^{p-1}}{d} - c\,(n\,(p-1)+1) \int \frac{(c \operatorname{ProductLog}[a x^n])^{p-1}}{d + d \operatorname{ProductLog}[a x^n]} \, dx$$

- **Program code:**

```
Int[(c_.*ProductLog[a_.*x_^n_.])^p_./(d_+d_.*ProductLog[a_.*x_^n_.]),x_Symbol] :=
  c*x*(c*ProductLog[a*x^n])^(p-1)/d -
  Dist[c*(n*(p-1)+1),Int[(c*ProductLog[a*x^n])^(p-1)/(d+d*ProductLog[a*x^n]),x]] /;
FreeQ[{a,c,d},x] && RationalQ[{n,p}] && n>0 && n*(p-1)+1>0
```

- **Rule: If** $n > 0 \bigwedge n\,p+1 < 0$, **then**

$$\int \frac{(c \operatorname{ProductLog}[a x^n])^p}{d + d \operatorname{ProductLog}[a x^n]} \, dx \longrightarrow \frac{x\, (c \operatorname{ProductLog}[a x^n])^{p}}{d\,(n\,p+1)} - \frac{1}{c\,(n\,p+1)} \int \frac{(c \operatorname{ProductLog}[a x^n])^{p+1}}{d + d \operatorname{ProductLog}[a x^n]} \, dx$$

- **Program code:**

```
Int[(c_.*ProductLog[a_.*x_^n_.])^p_./(d_+d_.*ProductLog[a_.*x_^n_.]),x_Symbol] :=
  x*(c*ProductLog[a*x^n])^p/(d*(n*p+1)) -
  Dist[1/(c*(n*p+1)),Int[(c*ProductLog[a*x^n])^(p+1)/(d+d*ProductLog[a*x^n]),x]] /;
FreeQ[{a,c,d},x] && RationalQ[{n,p}] && n>0 && n*p+1<0
```

- **Derivation: Integration by substitution**

- **Basis:** $\int f[x]\, dx = -\text{Subst}\left[\int \frac{f\left[\frac{1}{x}\right]}{x^2}\, dx,\ x,\ \frac{1}{x}\right]$

- **Rule: If** $n \in \mathbb{Z} \land n < 0$, **then**

$$\int \frac{(c\ \text{ProductLog}[a\,x^n])^p}{d + d\ \text{ProductLog}[a\,x^n]}\, dx \longrightarrow -\text{Subst}\left[\int \frac{(c\ \text{ProductLog}[a\,x^{-n}])^p}{x^2\ (d + d\ \text{ProductLog}[a\,x^{-n}])}\, dx,\ x,\ \frac{1}{x}\right]$$

- **Program code:**

```
Int[(c_.*ProductLog[a_.*x_^n_])^p_./(d_+d_.*ProductLog[a_.*x_^n_]),x_Symbol] :=
  -Subst[Int[(c*ProductLog[a*x^(-n)])^p/(x^2*(d+d*ProductLog[a*x^(-n)])),x],x,1/x] /;
FreeQ[{a,c,d,p},x] && IntegerQ[n] && n<0
```

$\int x^m\,(c\,\text{ProductLog}[a\,x^n])^p\,dx$

- **Program code:**

- **Derivation: Integration by parts**

- **Rule: If** $m+1\neq 0\,\bigwedge\,\left(p-\frac{1}{2}\in\mathbb{Z}\,\bigwedge\,2\left(p+\frac{m+1}{n}\right)\in\mathbb{Z}\,\bigwedge\,p+\frac{m+1}{n}>0\right)\,\bigvee\,\left(\neg\left(p-\frac{1}{2}\in\mathbb{Z}\right)\,\bigwedge\,p+\frac{m+1}{n}\in\mathbb{Z}\,\bigwedge\,p+\frac{m+1}{n}\geq 0\right)$**, then**

$$\int x^m\,(c\,\text{ProductLog}[a\,x^n])^p\,dx\;\longrightarrow$$
$$\frac{x^{m+1}\,(c\,\text{ProductLog}[a\,x^n])^p}{m+1}-\frac{n\,p}{m+1}\int\frac{x^m\,(c\,\text{ProductLog}[a\,x^n])^p}{1+\text{ProductLog}[a\,x^n]}\,dx$$

- **Program code:**

```
Int[x_^m_.*(c_.*ProductLog[a_.*x_^n_.])^p_.,x_Symbol] :=
  x^(m+1)*(c*ProductLog[a*x^n])^p/(m+1) -
  Dist[n*p/(m+1),Int[x^m*(c*ProductLog[a*x^n])^p/(1+ProductLog[a*x^n]),x]] /;
FreeQ[{a,c,m,n,p},x] && NonzeroQ[m+1] &&
(IntegerQ[p-1/2] && IntegerQ[2*Simplify[p+(m+1)/n]] && Simplify[p+(m+1)/n]>0 ||
 Not[IntegerQ[p-1/2]] && IntegerQ[Simplify[p+(m+1)/n]] && Simplify[p+(m+1)/n]>=0)
```

- **Rule: If** $m+1=0\,\bigvee\,\left(p-\frac{1}{2}\in\mathbb{Z}\,\bigwedge\,p+\frac{m+1}{n}-\frac{1}{2}\in\mathbb{Z}\,\bigwedge\,p+\frac{m+1}{n}<0\right)\,\bigvee\,\left(\neg\left(p-\frac{1}{2}\in\mathbb{Z}\right)\,\bigwedge\,p+\frac{m+1}{n}\in\mathbb{Z}\,\bigwedge\,p+\frac{m+1}{n}<0\right)$**, then**

$$\int x^m\,(c\,\text{ProductLog}[a\,x^n])^p\,dx\;\longrightarrow$$
$$\frac{x^{m+1}\,(c\,\text{ProductLog}[a\,x^n])^p}{m+n\,p+1}+\frac{n\,p}{c\,(m+n\,p+1)}\int\frac{x^m\,(c\,\text{ProductLog}[a\,x^n])^{p+1}}{1+\text{ProductLog}[a\,x^n]}\,dx$$

- **Program code:**

```
Int[x_^m_.*(c_.*ProductLog[a_.*x_^n_.])^p_.,x_Symbol] :=
  x^(m+1)*(c*ProductLog[a*x^n])^p/(m+n*p+1) +
  Dist[n*p/(c*(m+n*p+1)),Int[x^m*(c*ProductLog[a*x^n])^(p+1)/(1+ProductLog[a*x^n]),x]] /;
FreeQ[{a,c,m,n,p},x] &&
(ZeroQ[m+1] ||
 IntegerQ[p-1/2] && IntegerQ[Simplify[p+(m+1)/n]-1/2] && Simplify[p+(m+1)/n]<0 ||
 Not[IntegerQ[p-1/2]] && IntegerQ[Simplify[p+(m+1)/n]] && Simplify[p+(m+1)/n]<0)
```

- **Derivation: Algebraic simplification**
- **Basis:** $1 = \frac{1}{1+z} + \frac{z}{1+z}$
- **Rule:**

$$\int x^m \,(c\,\text{ProductLog}[a\,x])^p\,dx \longrightarrow \int \frac{x^m\,(c\,\text{ProductLog}[a\,x])^p}{1+\text{ProductLog}[a\,x]}\,dx + \frac{1}{c}\int \frac{x^m\,(c\,\text{ProductLog}[a\,x])^{p+1}}{1+\text{ProductLog}[a\,x]}\,dx$$

- **Program code:**

```
Int[x_^m_.*(c_.*ProductLog[a_.*x_])^p_.,x_Symbol] :=
  Int[x^m*(c*ProductLog[a*x])^p/(1+ProductLog[a*x]),x] +
  Dist[1/c,Int[x^m*(c*ProductLog[a*x])^(p+1)/(1+ProductLog[a*x]),x]] /;
FreeQ[{a,c,m},x]
```

- **Derivation: Integration by substitution**
- **Basis:** $\int f[x]\,dx = -\text{Subst}\left[\int \frac{f\left[\frac{1}{x}\right]}{x^2}\,dx,\ x,\ \frac{1}{x}\right]$
- **Rule: If** $m, n \in \mathbb{Z} \land n < 0 \land m+1 \neq 0$**, then**

$$\int x^m\,(c\,\text{ProductLog}[a\,x^n])^p\,dx \longrightarrow -\text{Subst}\left[\int \frac{(c\,\text{ProductLog}[a\,x^{-n}])^p}{x^{m+2}}\,dx,\ x,\ \frac{1}{x}\right]$$

- **Program code:**

```
Int[x_^m_.*(c_.*ProductLog[a_.*x_^n_])^p_.,x_Symbol] :=
  -Subst[Int[(c*ProductLog[a*x^(-n)])^p/x^(m+2),x],x,1/x] /;
FreeQ[{a,c,p},x] && IntegersQ[m,n] && n<0 && NonzeroQ[m+1]
```

$$\int \frac{x^m}{d + d\,\text{ProductLog}[a\,x^n]}\,dx$$

- **Rule:**

$$\int \frac{1}{x\,(d + d\,\text{ProductLog}[a\,x^n])}\,dx \longrightarrow \frac{\text{Log}[\text{ProductLog}[a\,x^n]]}{d\,n}$$

- **Program code:**

```
Int[1/(x_*(d_+d_.*ProductLog[a_.*x_^n_.])),x_Symbol] :=
  Log[ProductLog[a*x^n]]/(d*n) /;
FreeQ[{a,d,n},x]
```

- **Rule: If $m > 0$, then**

$$\int \frac{x^m}{d + d\,\text{ProductLog}[a\,x]}\,dx \longrightarrow$$
$$\frac{x^{m+1}}{d\,(m+1)\,\text{ProductLog}[a\,x]} - \frac{m}{m+1}\int \frac{x^m}{\text{ProductLog}[a\,x]\,(d + d\,\text{ProductLog}[a\,x])}\,dx$$

- **Program code:**

```
Int[x_^m_./(d_+d_.*ProductLog[a_.*x_]),x_Symbol] :=
  x^(m+1)/(d*(m+1)*ProductLog[a*x]) -
  Dist[m/(m+1),Int[x^m/(ProductLog[a*x]*(d+d*ProductLog[a*x])),x]] /;
FreeQ[{a,d},x] && RationalQ[m] && m>0
```

- **Rule: If $m < -1$, then**

$$\int \frac{x^m}{d + d\,\text{ProductLog}[a\,x]}\,dx \longrightarrow \frac{x^{m+1}}{d\,(m+1)} - \int \frac{x^m\,\text{ProductLog}[a\,x]}{d + d\,\text{ProductLog}[a\,x]}\,dx$$

- **Program code:**

```
Int[x_^m_./(d_+d_.*ProductLog[a_.*x_]),x_Symbol] :=
  x^(m+1)/(d*(m+1)) -
  Int[x^m*ProductLog[a*x]/(d+d*ProductLog[a*x]),x] /;
FreeQ[{a,d},x] && RationalQ[m] && m<-1
```

- **Rule: If** $m+1 \neq 0$**, then**

$$\int \frac{x^m}{d + d\,\text{ProductLog}[a\,x]}\,dx \longrightarrow \frac{x^m\,\text{Gamma}[m+1, -(m+1)\,\text{ProductLog}[a\,x]]}{a\,d\,(m+1)\,e^{m\,\text{ProductLog}[a\,x]}\,(-(m+1)\,\text{ProductLog}[a\,x])^m}$$

- **Program code:**

```
Int[x_^m_./(d_+d_.*ProductLog[a_.*x_]),x_Symbol] :=
  x^m*Gamma[m+1,-(m+1)*ProductLog[a*x]]/
    (a*d*(m+1)*E^(m*ProductLog[a*x])*(-(m+1)*ProductLog[a*x])^m) /;
FreeQ[{a,d},x] && NonzeroQ[m+1]
```

- **Derivation: Integration by substitution**
- **Basis:** $\int f[x]\,dx = -\text{Subst}\left[\int \frac{f\left[\frac{1}{x}\right]}{x^2}\,dx, x, \frac{1}{x}\right]$
- **Rule: If** $m, n \in \mathbb{Z} \wedge n < 0 \wedge m+1 \neq 0$**, then**

$$\int \frac{x^m}{d + d\,\text{ProductLog}[a\,x^n]}\,dx \longrightarrow -\text{Subst}\left[\int \frac{1}{x^{m+2}\,(d + d\,\text{ProductLog}[a\,x^{-n}])}\,dx, x, \frac{1}{x}\right]$$

- **Program code:**

```
Int[x_^m_./(d_+d_.*ProductLog[a_.*x_^n_]),x_Symbol] :=
  -Subst[Int[1/(x^(m+2)*(d+d*ProductLog[a*x^(-n)])),x],x,1/x] /;
FreeQ[{a,d},x] && IntegersQ[m,n] && n<0 && NonzeroQ[m+1]
```

$$\int \frac{x^m\,(c\,\text{ProductLog}[a\,x^n])^p}{d+d\,\text{ProductLog}[a\,x^n]}\,dx$$

- **Rule:**

$$\int \frac{(c\,\text{ProductLog}[a\,x^n])^p}{x\,(d+d\,\text{ProductLog}[a\,x^n])}\,dx \longrightarrow \frac{(c\,\text{ProductLog}[a\,x^n])^p}{d\,n\,p}$$

- **Program code:**

```
Int[(c_.*ProductLog[a_.*x_^n_.])^p_./(x_*(d_+d_.*ProductLog[a_.*x_^n_.])),x_Symbol] :=
  (c*ProductLog[a*x^n])^p/(d*n*p) /;
FreeQ[{a,c,d,n,p},x]
```

- **Rule: If** $m+n\,(p-1)+1=0\,\wedge\,m+1\neq 0$**, then**

$$\int \frac{x^m\,(c\,\text{ProductLog}[a\,x^n])^p}{d+d\,\text{ProductLog}[a\,x^n]}\,dx \longrightarrow \frac{c\,x^{m+1}\,(c\,\text{ProductLog}[a\,x^n])^{p-1}}{d\,(m+1)}$$

- **Program code:**

```
Int[x_^m_.*(c_.*ProductLog[a_.*x_^n_.])^p_./(d_+d_.*ProductLog[a_.*x_^n_.]),x_Symbol] :=
  c*x^(m+1)*(c*ProductLog[a*x^n])^(p-1)/(d*(m+1)) /;
FreeQ[{a,c,d,m,n,p},x] && ZeroQ[m+n*(p-1)+1] && NonzeroQ[m+1]
```

- **Rule: If** $p\in\mathbb{Z}\,\wedge\,m+n\,p+1=0$**, then**

$$\int \frac{x^m\,\text{ProductLog}[a\,x^n]^p}{d+d\,\text{ProductLog}[a\,x^n]}\,dx \longrightarrow \frac{a^p\,\text{ExpIntegralEi}[-p\,\text{ProductLog}[a\,x^n]]}{d\,n}$$

- **Program code:**

```
Int[x_^m_.*ProductLog[a_.*x_^n_.]^p_./(d_+d_.*ProductLog[a_.*x_^n_.]),x_Symbol] :=
  a^p*ExpIntegralEi[-p*ProductLog[a*x^n]]/(d*n) /;
FreeQ[{a,d,m,n},x] && IntegerQ[p] && ZeroQ[m+n*p+1]
```

- **Rule: If** $p-\frac{1}{2}\in\mathbb{Z}\bigwedge p-\frac{1}{2}\neq 0\bigwedge m+n\left(p-\frac{1}{2}\right)+1=0\bigwedge \frac{c}{p-\frac{1}{2}}>0$**, then**

$$\int\frac{x^m\,(c\,\text{ProductLog}[a\,x^n])^p}{d+d\,\text{ProductLog}[a\,x^n]}\,dx\;\longrightarrow\;\frac{a^{p-\frac{1}{2}}\,c^{p-\frac{1}{2}}}{d\,n}\sqrt{\frac{\pi\,c}{p-\frac{1}{2}}}\;\text{Erf}\left[\frac{\sqrt{c\,\text{ProductLog}[a\,x^n]}}{\sqrt{\frac{c}{p-\frac{1}{2}}}}\right]$$

- **Program code:**

```
Int[x_^m_.*(c_.*ProductLog[a_.*x_^n_.])^p_/(d_+d_.*ProductLog[a_.*x_^n_.]),x_Symbol] :=
  a^(p-1/2)*c^(p-1/2)*Rt[Pi*c/(p-1/2),2]*Erf[Sqrt[c*ProductLog[a*x^n]]/Rt[c/(p-1/2),2]]/(d*n) /;
FreeQ[{a,c,d,m,n},x] && IntegerQ[p-1/2] && p-1/2≠0 && ZeroQ[m+n*(p-1/2)+1] && PosQ[c/(p-1/2)]
```

- **Rule: If** $p-\frac{1}{2}\in\mathbb{Z}\bigwedge p-\frac{1}{2}\neq 0\bigwedge m+n\left(p-\frac{1}{2}\right)+1=0\bigwedge \frac{c}{p-\frac{1}{2}}<0$**, then**

$$\int\frac{x^m\,(c\,\text{ProductLog}[a\,x^n])^p}{d+d\,\text{ProductLog}[a\,x^n]}\,dx\;\longrightarrow\;\frac{a^{p-\frac{1}{2}}\,c^{p-\frac{1}{2}}}{d\,n}\sqrt{-\frac{\pi\,c}{p-\frac{1}{2}}}\;\text{Erfi}\left[\frac{\sqrt{c\,\text{ProductLog}[a\,x^n]}}{\sqrt{-\frac{c}{p-\frac{1}{2}}}}\right]$$

- **Program code:**

```
Int[x_^m_.*(c_.*ProductLog[a_.*x_^n_.])^p_/(d_+d_.*ProductLog[a_.*x_^n_.]),x_Symbol] :=
  a^(p-1/2)*c^(p-1/2)*Rt[-Pi*c/(p-1/2),2]*Erfi[Sqrt[c*ProductLog[a*x^n]]/Rt[-c/(p-1/2),2]]/(d*n) /;
FreeQ[{a,c,d,m,n},x] && IntegerQ[p-1/2] && p-1/2≠0 && ZeroQ[m+n*(p-1/2)+1] && NegQ[c/(p-1/2)]
```

- **Rule: If** $m+1\neq 0\bigwedge p+\frac{m+1}{n}>1$**, then**

$$\int\frac{x^m\,(c\,\text{ProductLog}[a\,x^n])^p}{d+d\,\text{ProductLog}[a\,x^n]}\,dx\;\longrightarrow$$
$$\frac{c\,x^{m+1}\,(c\,\text{ProductLog}[a\,x^n])^{p-1}}{d\,(m+1)}-\frac{c\,(m+n\,(p-1)+1)}{m+1}\int\frac{x^m\,(c\,\text{ProductLog}[a\,x^n])^{p-1}}{d+d\,\text{ProductLog}[a\,x^n]}\,dx$$

- **Program code:**

```
Int[x_^m_.*(c_.*ProductLog[a_.*x_^n_.])^p_./(d_+d_.*ProductLog[a_.*x_^n_.]),x_Symbol] :=
  c*x^(m+1)*(c*ProductLog[a*x^n])^(p-1)/(d*(m+1)) -
  Dist[c*(m+n*(p-1)+1)/(m+1),Int[x^m*(c*ProductLog[a*x^n])^(p-1)/(d+d*ProductLog[a*x^n]),x]] /;
FreeQ[{a,c,d,m,n,p},x] && NonzeroQ[m+1] && RationalQ[Simplify[p+(m+1)/n]] && Simplify[p+(m+1)/n]>1
```

- **Rule: If** $m+1\neq 0\bigwedge p+\frac{m+1}{n}<0$**, then**

$$\int\frac{x^m\,(c\,\text{ProductLog}[a\,x^n])^p}{d+d\,\text{ProductLog}[a\,x^n]}\,dx\longrightarrow$$
$$\frac{x^{m+1}\,(c\,\text{ProductLog}[a\,x^n])^p}{d\,(m+n\,p+1)}-\frac{m+1}{c\,(m+n\,p+1)}\int\frac{x^m\,(c\,\text{ProductLog}[a\,x^n])^{p+1}}{d+d\,\text{ProductLog}[a\,x^n]}\,dx$$

- **Program code:**

```
Int[x_^m_.*(c_.*ProductLog[a_.*x_^n_.])^p_./(d_+d_.*ProductLog[a_.*x_^n_.]),x_Symbol] :=
  x^(m+1)*(c*ProductLog[a*x^n])^p/(d*(m+n*p+1)) -
  Dist[(m+1)/(c*(m+n*p+1)),Int[x^m*(c*ProductLog[a*x^n])^(p+1)/(d+d*ProductLog[a*x^n]),x]] /;
FreeQ[{a,c,d,m,n,p},x] && NonzeroQ[m+1] && RationalQ[Simplify[p+(m+1)/n]] && Simplify[p+(m+1)/n]<0
```

- **Rule: If** $m+1\neq 0$**, then**

$$\int\frac{x^m\,(c\,\text{ProductLog}[a\,x])^p}{d+d\,\text{ProductLog}[a\,x]}\,dx\longrightarrow\frac{x^m\,\text{Gamma}[m+p+1,\,-(m+1)\,\text{ProductLog}[a\,x]]\,(c\,\text{ProductLog}[a\,x])^p}{a\,d\,(m+1)\,e^{m\,\text{ProductLog}[a\,x]}\,(-(m+1)\,\text{ProductLog}[a\,x])^{m+p}}$$

- **Program code:**

```
Int[x_^m_.*(c_.*ProductLog[a_.*x_])^p_./(d_+d_.*ProductLog[a_.*x_]),x_Symbol] :=
  x^m*Gamma[m+p+1,-(m+1)*ProductLog[a*x]]*(c*ProductLog[a*x])^p/
    (a*d*(m+1)*E^(m*ProductLog[a*x])*(-(m+1)*ProductLog[a*x])^(m+p)) /;
FreeQ[{a,c,d,m,p},x] && NonzeroQ[m+1]
```

- **Derivation: Integration by substitution**

- **Basis:** $\int f[x]\,dx=-\text{Subst}\left[\int\frac{f\left[\frac{1}{x}\right]}{x^2}\,dx,\,x,\,\frac{1}{x}\right]$

- **Rule: If** $m,\,n\in\mathbb{Z}\bigwedge n<0\bigwedge m+1\neq 0$**, then**

$$\int\frac{x^m\,(c\,\text{ProductLog}[a\,x^n])^p}{d+d\,\text{ProductLog}[a\,x^n]}\,dx\longrightarrow-\text{Subst}\left[\int\frac{(c\,\text{ProductLog}[a\,x^{-n}])^p}{x^{m+2}\,(d+d\,\text{ProductLog}[a\,x^{-n}])}\,dx,\,x,\,\frac{1}{x}\right]$$

- **Program code:**

```
Int[x_^m_.*(c_.*ProductLog[a_.*x_^n_.])^p_./(d_+d_.*ProductLog[a_.*x_^n_.]),x_Symbol] :=
  -Subst[Int[(c*ProductLog[a*x^(-n)])^p/(x^(m+2)*(d+d*ProductLog[a*x^(-n)])),x],x,1/x] /;
FreeQ[{a,c,d,p},x] && IntegersQ[m,n] && n<0 && NonzeroQ[m+1]
```

$$\int f[\text{ProductLog}[a + b\,x]]\,dx$$

- **Author: Rob Corless 2009-07-10**
- **Derivation: Legendre substitution for inverse functions**
- **Basis:** $f[\text{ProductLog}[x]] = (\text{ProductLog}[z] + 1)\, e^{\text{ProductLog}[z]}\, f[\text{ProductLog}[x]]\, \text{ProductLog}'[z]$
- **Rule:**

$$\int f[\text{ProductLog}[x]]\,dx \longrightarrow \text{Subst}\left[\int (x+1)\, e^{x}\, f[x]\,dx,\, x,\, \text{ProductLog}[x]\right]$$

- **Program code:**

```
Int[u_,x_Symbol] :=
  Subst[Int[Regularize[(x+1)*E^x*SubstFor[ProductLog[x],u,x],x],x],x,ProductLog[x]] /;
FunctionOfQ[ProductLog[x],u,x]
```