$\int \text{ArcSin}[a + b\,x]^n\, dx$

- **Reference: G&R 2.813.1, CRC 441, A&S 4.4.58**
- **Derivation: Integration by parts**
- **Rule:**

$$\int \text{ArcSin}[a + b\,x]\, dx \longrightarrow \frac{(a + b\,x)\ \text{ArcSin}[a + b\,x]}{b} + \frac{\sqrt{1 - (a + b\,x)^2}}{b}$$

- **Program code:**

```
Int[ArcSin[a_.+b_.*x_],x_Symbol] :=
  (a+b*x)*ArcSin[a+b*x]/b + Sqrt[1-(a+b*x)^2]/b /;
FreeQ[{a,b},x]
```

- **Reference: CRC 465**
- **Derivation: Iterated integration by parts**
- **Rule: If $n > 1$, then**

$$\int \text{ArcSin}[a + b\,x]^n\, dx \longrightarrow \frac{(a + b\,x)\ \text{ArcSin}[a + b\,x]^n}{b} + \frac{n\,\sqrt{1 - (a + b\,x)^2}\ \text{ArcSin}[a + b\,x]^{n-1}}{b} - n\,(n - 1) \int \text{ArcSin}[a + b\,x]^{n-2}\, dx$$

- **Program code:**

```
Int[ArcSin[a_.+b_.*x_]^n_,x_Symbol] :=
  (a+b*x)*ArcSin[a+b*x]^n/b +
  n*Sqrt[1-(a+b*x)^2]*ArcSin[a+b*x]^(n-1)/b -
  Dist[n*(n-1),Int[ArcSin[a+b*x]^(n-2),x]] /;
FreeQ[{a,b},x] && RationalQ[n] && n>1
```

- **Derivation: Integration by substitution**
- **Basis:** $\frac{1}{\text{ArcSin}[z]} = \frac{\text{Cos}[\text{ArcSin}[z]]}{\text{ArcSin}[z]} \text{ArcSin}'[z]$
- **Rule:**

$$\int \frac{1}{\text{ArcSin}[a+b\,x]}\,dx \longrightarrow \frac{\text{CosIntegral}[\text{ArcSin}[a+b\,x]]}{b}$$

- **Program code:**

```
Int[1/ArcSin[a_.+b_.*x_],x_Symbol] :=
  CosIntegral[ArcSin[a+b*x]]/b /;
FreeQ[{a,b},x]
```

- **Derivation: Integration by substitution**
- **Basis:** $\frac{1}{\text{ArcSin}[z]^2} = \frac{\text{Cos}[\text{ArcSin}[z]]}{\text{ArcSin}[z]^2} \text{ArcSin}'[z]$
- **Rule:**

$$\int \frac{1}{\text{ArcSin}[a+b\,x]^2}\,dx \longrightarrow -\frac{\sqrt{1-(a+b\,x)^2}}{b\,\text{ArcSin}[a+b\,x]} - \frac{\text{SinIntegral}[\text{ArcSin}[a+b\,x]]}{b}$$

- **Program code:**

```
Int[1/ArcSin[a_.+b_.*x_]^2,x_Symbol] :=
  -Sqrt[1-(a+b*x)^2]/(b*ArcSin[a+b*x]) - SinIntegral[ArcSin[a+b*x]]/b /;
FreeQ[{a,b},x]
```

- **Derivation: Integration by substitution**
- **Basis:** $\frac{1}{\sqrt{\text{ArcSin}[z]}} = \frac{\text{Cos}[\text{ArcSin}[z]]}{\sqrt{\text{ArcSin}[z]}} \text{ArcSin}'[z]$
- **Rule:**

$$\int \frac{1}{\sqrt{\text{ArcSin}[a+b\,x]}}\,dx \longrightarrow \frac{1}{b}\sqrt{2\,\pi}\ \text{FresnelC}\left[\sqrt{\frac{2}{\pi}}\ \sqrt{\text{ArcSin}[a+b\,x]}\right]$$

- **Program code:**

```
Int[1/Sqrt[ArcSin[a_.+b_.*x_]],x_Symbol] :=
  Sqrt[2*Pi]*FresnelC[Sqrt[2/Pi]*Sqrt[ArcSin[a+b*x]]]/b /;
FreeQ[{a,b},x]
```

- **Derivation: Integration by parts**
- **Rule:**

$$\int\sqrt{\text{ArcSin}[a+b\,x]}\,dx\;\longrightarrow\;\frac{(a+b\,x)\sqrt{\text{ArcSin}[a+b\,x]}}{b}-\frac{1}{b}\sqrt{\frac{\pi}{2}}\;\text{FresnelS}\left[\sqrt{\frac{2}{\pi}}\;\sqrt{\text{ArcSin}[a+b\,x]}\right]$$

- **Program code:**

```
Int[Sqrt[ArcSin[a_.+b_.*x_]],x_Symbol] :=
  (a+b*x)*Sqrt[ArcSin[a+b*x]]/b -
  Sqrt[Pi/2]*FresnelS[Sqrt[2/Pi]*Sqrt[ArcSin[a+b*x]]]/b /;
FreeQ[{a,b},x]
```

- **Derivation: Inverted iterated integration by parts**
- **Rule:** If $n<-1\land n\neq-2$, then

$$\int\text{ArcSin}[a+b\,x]^n\,dx\;\longrightarrow\;\frac{(a+b\,x)\,\text{ArcSin}[a+b\,x]^{n+2}}{b\,(n+1)\,(n+2)}+\frac{\sqrt{1-(a+b\,x)^2}\;\text{ArcSin}[a+b\,x]^{n+1}}{b\,(n+1)}-\frac{1}{(n+1)\,(n+2)}\int\text{ArcSin}[a+b\,x]^{n+2}\,dx$$

- **Program code:**

```
Int[ArcSin[a_.+b_.*x_]^n_,x_Symbol] :=
  (a+b*x)*ArcSin[a+b*x]^(n+2)/(b*(n+1)*(n+2)) +
  Sqrt[1-(a+b*x)^2]*ArcSin[a+b*x]^(n+1)/(b*(n+1)) -
  Dist[1/((n+1)*(n+2)),Int[ArcSin[a+b*x]^(n+2),x]] /;
FreeQ[{a,b},x] && RationalQ[n] && n<-1 && n!=-2
```

- **Rule:** If $n\notin\mathbb{Q}\lor-1<n<1$, then

$$\int\text{ArcSin}[a+b\,x]^n\,dx\;\longrightarrow\;\frac{i\,\text{ArcSin}[a+b\,x]^n}{2\,b}\left(\frac{\Gamma[n+1,\,i\,\text{ArcSin}[a+b\,x]]}{(i\,\text{ArcSin}[a+b\,x])^n}-\frac{\Gamma[n+1,\,-i\,\text{ArcSin}[a+b\,x]]}{(-i\,\text{ArcSin}[a+b\,x])^n}\right)$$

- **Program code:**

```
Int[ArcSin[a_.+b_.*x_]^n_,x_Symbol] :=
  I*ArcSin[a+b*x]^n/(2*b)*
    (Gamma[n+1,I*ArcSin[a+b*x]]/(I*ArcSin[a+b*x])^n -
     Gamma[n+1,-I*ArcSin[a+b*x]]/(-I*ArcSin[a+b*x])^n) /;
FreeQ[{a,b,n},x] && (Not[RationalQ[n]] || -1<n<1)
```

$$\int x^m \, \mathrm{ArcSin}[a + b\,x] \, dx$$

- **Reference: G&R 2.831, CRC 453, A&S 4.4.65**
- **Derivation: Integration by parts**
- **Rule: If $m + 1 \neq 0$, then**

$$\int x^m \, \mathrm{ArcSin}[a + b\,x] \, dx \longrightarrow \frac{x^{m+1} \, \mathrm{ArcSin}[a + b\,x]}{m + 1} - \frac{b}{m + 1} \int \frac{x^{m+1}}{\sqrt{1 - a^2 - 2\,a\,b\,x - b^2\,x^2}} \, dx$$

- **Program code:**

```
Int[x_^m_.*ArcSin[a_.+b_.*x_],x_Symbol] :=
  x^(m+1)*ArcSin[a+b*x]/(m+1) -
  Dist[b/(m+1),Int[x^(m+1)/Sqrt[1-a^2-2*a*b*x-b^2*x^2],x]] /;
FreeQ[{a,b,m},x] && NonzeroQ[m+1]
```

$$\int x^m \,\text{ArcSin}[a\,x]^n\, dx$$

- **Rule:**

$$\int \frac{x}{\sqrt{\text{ArcSin}[a\,x]}}\, dx \longrightarrow \frac{\sqrt{\pi}}{2\,a^2}\,\text{FresnelS}\Big[\frac{2\sqrt{\text{ArcSin}[a\,x]}}{\sqrt{\pi}}\Big]$$

- **Program code:**

```
Int[x_/Sqrt[ArcSin[a_.*x_]],x_Symbol] :=
  Sqrt[Pi]/(2*a^2)*FresnelS[2*Sqrt[ArcSin[a*x]]/Sqrt[Pi]] /;
FreeQ[a,x]
```

- **Rule:**

$$\int \frac{x}{\text{ArcSin}[a\,x]^{3/2}}\, dx \longrightarrow -\frac{2\,x\sqrt{1-a^2\,x^2}}{a\sqrt{\text{ArcSin}[a\,x]}} + \frac{2\sqrt{\pi}}{a^2}\,\text{FresnelC}\Big[\frac{2\sqrt{\text{ArcSin}[a\,x]}}{\sqrt{\pi}}\Big]$$

- **Program code:**

```
Int[x_/ArcSin[a_.*x_]^(3/2),x_Symbol] :=
  -2*x*Sqrt[1-a^2*x^2]/(a*Sqrt[ArcSin[a*x]]) + 2*Sqrt[Pi]/a^2*FresnelC[2*Sqrt[ArcSin[a*x]]/Sqrt[Pi]]
FreeQ[a,x]
```

- **Rule: If $n > 1$, then**

$$\int x\,\text{ArcSin}[a\,x]^n\, dx \longrightarrow \frac{n\,x\sqrt{1-a^2\,x^2}\;\text{ArcSin}[a\,x]^{n-1}}{4\,a} - \frac{\text{ArcSin}[a\,x]^n}{4\,a^2} + \frac{x^2\,\text{ArcSin}[a\,x]^n}{2} - \frac{n\,(n-1)}{4}\int x\,\text{ArcSin}[a\,x]^{n-2}\, dx$$

- **Program code:**

```
Int[x_*ArcSin[a_.*x_]^n_,x_Symbol] :=
  n*x*Sqrt[1-a^2*x^2]*ArcSin[a*x]^(n-1)/(4*a) -
  ArcSin[a*x]^n/(4*a^2) + x^2*ArcSin[a*x]^n/2 -
  Dist[n*(n-1)/4,Int[x*ArcSin[a*x]^(n-2),x]] /;
FreeQ[a,x] && RationalQ[n] && n>0
```

- **Rule: If** $n < -1 \land n \neq -2$, **then**

$$\int x \operatorname{ArcSin}[a x]^n \, dx \longrightarrow \frac{x \sqrt{1-a^2 x^2} \operatorname{ArcSin}[a x]^{n+1}}{a (n+1)} - \frac{\operatorname{ArcSin}[a x]^{n+2}}{a^2 (n+1) (n+2)} + \frac{2 x^2 \operatorname{ArcSin}[a x]^{n+2}}{(n+1) (n+2)} - \frac{4}{(n+1) (n+2)} \int x \operatorname{ArcSin}[a x]^{n+2} \, dx$$

- **Program code:**

```
Int[x_*ArcSin[a_.*x_]^n_,x_Symbol] :=
  x*Sqrt[1-a^2*x^2]*ArcSin[a*x]^(n+1)/(a*(n+1)) -
  ArcSin[a*x]^(n+2)/(a^2*(n+1)*(n+2)) +
  2*x^2*ArcSin[a*x]^(n+2)/((n+1)*(n+2)) -
  Dist[4/((n+1)*(n+2)),Int[x*ArcSin[a*x]^(n+2),x]] /;
FreeQ[a,x] && RationalQ[n] && n<-1 && n≠-2
```

- **Rule: If** $n > 1$, **then**

$$\int \frac{\operatorname{ArcSin}[a x]^n}{x^3} \, dx \longrightarrow -\frac{a n \sqrt{1-a^2 x^2} \operatorname{ArcSin}[a x]^{n-1}}{2 x} - \frac{\operatorname{ArcSin}[a x]^n}{2 x^2} + \frac{a^2 n (n-1)}{2} \int \frac{\operatorname{ArcSin}[a x]^{n-2}}{x} \, dx$$

- **Program code:**

```
Int[ArcSin[a_.*x_]^n_/x_^3,x_Symbol] :=
  -a*n*Sqrt[1-a^2*x^2]*ArcSin[a*x]^(n-1)/(2*x) -
   ArcSin[a*x]^n/(2*x^2) +
  Dist[a^2*n*(n-1)/2,Int[ArcSin[a*x]^(n-2)/x,x]] /;
FreeQ[a,x] && RationalQ[n] && n>1
```

- **Rule: If** $m \in \mathbb{Z} \wedge m < -3 \wedge n > 1$, **then**

$$\int x^m \operatorname{ArcSin}[a\,x]^n \, dx \longrightarrow -\frac{a\,n\,x^{m+2}\sqrt{1-a^2x^2}\,\operatorname{ArcSin}[a\,x]^{n-1}}{(m+1)\,(m+2)} + \frac{x^{m+1}\operatorname{ArcSin}[a\,x]^n}{(m+1)} - \frac{a^2\,(m+3)\,x^{m+3}\operatorname{ArcSin}[a\,x]^n}{(m+1)\,(m+2)} + \frac{a^2\,(m+3)^2}{(m+1)\,(m+2)}\int x^{m+2}\operatorname{ArcSin}[a\,x]^n\,dx + \frac{a^2\,n\,(n-1)}{(m+1)\,(m+2)}\int x^{m+2}\operatorname{ArcSin}[a\,x]^{n-2}\,dx$$

- **Program code:**

```
Int[x_^m_*ArcSin[a_.*x_]^n_,x_Symbol] :=
  -a*n*x^(m+2)*Sqrt[1-a^2*x^2]*ArcSin[a*x]^(n-1)/((m+1)*(m+2)) +
  x^(m+1)*ArcSin[a*x]^n/(m+1) -
  a^2*(m+3)*x^(m+3)*ArcSin[a*x]^n/((m+1)*(m+2)) +
  Dist[a^2*(m+3)^2/((m+1)*(m+2)),Int[x^(m+2)*ArcSin[a*x]^n,x]] +
  Dist[a^2*n*(n-1)/((m+1)*(m+2)),Int[x^(m+2)*ArcSin[a*x]^(n-2),x]] /;
FreeQ[a,x] && IntegerQ[m] && RationalQ[n] && m<-3 && n>1
```

- **Rule: If** $m \in \mathbb{Z} \wedge m > 1 \wedge n < -1 \wedge n \neq -2$, **then**

$$\int x^m \operatorname{ArcSin}[a\,x]^n \, dx \longrightarrow \frac{x^m\sqrt{1-a^2x^2}\,\operatorname{ArcSin}[a\,x]^{n+1}}{a\,(n+1)} - \frac{m\,x^{m-1}\operatorname{ArcSin}[a\,x]^{n+2}}{a^2\,(n+1)\,(n+2)} + \frac{(m+1)\,x^{m+1}\operatorname{ArcSin}[a\,x]^{n+2}}{(n+1)\,(n+2)} - \frac{(m+1)^2}{(n+1)\,(n+2)}\int x^m\operatorname{ArcSin}[a\,x]^{n+2}\,dx + \frac{m\,(m-1)}{a^2\,(n+1)\,(n+2)}\int x^{m-2}\operatorname{ArcSin}[a\,x]^{n+2}\,dx$$

- **Program code:**

```
Int[x_^m_*ArcSin[a_.*x_]^n_,x_Symbol] :=
  x^m*Sqrt[1-a^2*x^2]*ArcSin[a*x]^(n+1)/(a*(n+1)) -
  m*x^(m-1)*ArcSin[a*x]^(n+2)/(a^2*(n+1)*(n+2)) +
  (m+1)*x^(m+1)*ArcSin[a*x]^(n+2)/((n+1)*(n+2)) -
  Dist[(m+1)^2/((n+1)*(n+2)),Int[x^m*ArcSin[a*x]^(n+2),x]] +
  Dist[m*(m-1)/(a^2*(n+1)*(n+2)),Int[x^(m-2)*ArcSin[a*x]^(n+2),x]] /;
FreeQ[a,x] && IntegerQ[m] && RationalQ[n] && m>1 && n<-1 && n≠-2
```

- **Derivation: Integration by substitution**
- **Basis:** $\frac{\text{ArcSin}[a\,x^p]^n}{x} = \frac{1}{p}\,\text{ArcSin}[a\,x^p]^n\,\text{Cot}[\text{ArcSin}[a\,x^p]]\,\partial_x\text{ArcSin}[a\,x^p]$
- **Rule: If** $n \in \mathbb{Z} \wedge n > 0$**, then**

$$\int \frac{\text{ArcSin}[a\,x^p]^n}{x}\,dx \longrightarrow \frac{1}{p}\,\text{Subst}\left[\int x^n\,\text{Cot}[x]\,dx,\,x,\,\text{ArcSin}[a\,x^p]\right]$$

- **Program code:**

```
Int[ArcSin[a_.*x_^p_.]^n_./x_,x_Symbol] :=
  Dist[1/p,Subst[Int[x^n*Cot[x],x],x,ArcSin[a*x^p]]] /;
FreeQ[{a,p},x] && IntegerQ[n] && n>0
```

- **Derivation: Integration by parts and substitution**
- **Basis: If** $m \in \mathbb{Z}$**,** $\frac{x^{m+1}\,\text{ArcSin}[a\,x]^{n-1}}{\sqrt{1-a^2\,x^2}} = \frac{1}{a^{m+2}}\,\text{ArcSin}[a\,x]^{n-1}\,\text{Sin}[\text{ArcSin}[a\,x]]^{m+1}\,\partial_x\text{ArcSin}[a\,x]$
- **Rule: If** $m \in \mathbb{Z} \wedge m \neq -1$**, then**

$$\int x^m\,\text{ArcSin}[a\,x]^n\,dx \longrightarrow \frac{x^{m+1}\,\text{ArcSin}[a\,x]^n}{m+1} - \frac{n}{a^{m+1}\,(m+1)}\,\text{Subst}\left[\int x^{n-1}\,\text{Sin}[x]^{m+1}\,dx,\,x,\,\text{ArcSin}[a\,x]\right]$$

- **Program code:**

```
Int[x_^m_.*ArcSin[a_.*x_]^n_,x_Symbol] :=
  x^(m+1)*ArcSin[a*x]^n/(m+1) -
  Dist[n/(a^(m+1)*(m+1)),Subst[Int[x^(n-1)*Sin[x]^(m+1),x],x,ArcSin[a*x]]] /;
FreeQ[{a,n},x] && IntegerQ[m] && m≠-1
```

$$\int (a + b \operatorname{ArcSin}[c + d x])^n \, dx$$

- **Derivation: Integration by substitution**

- **Basis:** $(a + b \operatorname{ArcSin}[c + d x])^n = \frac{1}{d} (a + b \operatorname{ArcSin}[c + d x])^n \operatorname{Cos}[\operatorname{ArcSin}[c + d x]] \, \partial_x \operatorname{ArcSin}[c + d x]$

- **Rule: If** $n \notin \mathbb{Z}$**, then**

$$\int (a + b \operatorname{ArcSin}[c + d x])^n \, dx \longrightarrow \frac{1}{d} \operatorname{Subst}\left[\int (a + b x)^n \operatorname{Cos}[x] \, dx, x, \operatorname{ArcSin}[c + d x]\right]$$

- **Program code:**

```
Int[(a_+b_.*ArcSin[c_.+d_.*x_])^n_,x_Symbol] :=
  Dist[1/d,Subst[Int[(a+b*x)^n*Cos[x],x],x,ArcSin[c+d*x]]] /;
FreeQ[{a,b,c,d},x] && Not[IntegerQ[n]]
```

$$\int x^m\,(a+b\,\mathrm{ArcSin}[c+d\,x])^n\,dx$$

- **Derivation: Integration by substitution**

- **Basis: If** $m\in\mathbb{Z}$, $x^m\,(a+b\,\mathrm{ArcSin}[c+d\,x])^n=$
$$\frac{1}{d^{m+1}}\,(a+b\,\mathrm{ArcSin}[c+d\,x])^n\,(\mathrm{Sin}[\mathrm{ArcSin}[c+d\,x]]-c)^m\,\mathrm{Cos}[\mathrm{ArcSin}[c+d\,x]]\,\partial_x\mathrm{ArcSin}[c+d\,x]$$

- **Rule: If** $m\in\mathbb{Z}\wedge n\notin\mathbb{Z}\wedge m>0$**, then**

$$\int x^m\,(a+b\,\mathrm{ArcSin}[c+d\,x])^n\,dx\;\longrightarrow\;\frac{1}{d^{m+1}}\,\mathrm{Subst}\Big[\int(a+b\,x)^n\,(\mathrm{Sin}[x]-c)^m\,\mathrm{Cos}[x]\,dx,\;x,\;\mathrm{ArcSin}[c+d\,x]\Big]$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*ArcSin[c_.+d_.*x_])^n_,x_Symbol] :=
  Dist[1/d^(m+1),Subst[Int[(a+b*x)^n*(Sin[x]-c)^m*Cos[x],x],x,ArcSin[c+d*x]]] /;
FreeQ[{a,b,c,d},x] && IntegerQ[m] && Not[IntegerQ[n]] && m>0
```

$$\int \frac{x \operatorname{ArcSin}[a+b x]^n}{\sqrt{1-(a+b x)^2}} \, dx$$

- **Derivation: Integration by parts**
- **Rule: If $n > 1$, then**

$$\int \frac{x \operatorname{ArcSin}[a+b x]^n}{\sqrt{1-(a+b x)^2}} \, dx \longrightarrow$$

$$-\frac{\sqrt{1-(a+b x)^2} \operatorname{ArcSin}[a+b x]^n}{b^2}+\frac{n}{b} \int \operatorname{ArcSin}[a+b x]^{n-1} \, dx-\frac{a}{b} \int \frac{\operatorname{ArcSin}[a+b x]^n}{\sqrt{1-(a+b x)^2}} \, dx$$

- **Program code:**

```
Int[x_*ArcSin[a_.+b_.*x_]^n_/Sqrt[u_],x_Symbol] :=
  -Sqrt[u]*ArcSin[a+b*x]^n/b^2 +
  Dist[n/b,Int[ArcSin[a+b*x]^(n-1),x]] -
  Dist[a/b,Int[ArcSin[a+b*x]^n/Sqrt[u],x]] /;
FreeQ[{a,b},x] && ZeroQ[u-1+(a+b*x)^2] && RationalQ[n] && n>1
```

$$\int u \operatorname{ArcSin}\left[\frac{c}{a + b x^n}\right]^m dx$$

- **Derivation: Algebraic simplification**
- **Basis:** $\operatorname{ArcSin}[z] = \operatorname{ArcCsc}\left[\frac{1}{z}\right]$
- **Rule:**

$$\int u \operatorname{ArcSin}\left[\frac{c}{a + b x^n}\right]^m dx \longrightarrow \int u \operatorname{ArcCsc}\left[\frac{a}{c} + \frac{b x^n}{c}\right]^m dx$$

- **Program code:**

```
Int[u_.*ArcSin[c_./(a_.+b_.*x_^n_.)]^m_.,x_Symbol] :=
  Int[u*ArcCsc[a/c+b*x^n/c]^m,x] /;
FreeQ[{a,b,c,n,m},x]
```

$$\int f^{c\,\text{ArcSin}[a+b\,x]}\,dx$$

- **Rule: If** $1+c^2\,\text{Log}[f]^2\neq 0$**, then**

$$\int f^{c\,\text{ArcSin}[a+b\,x]}\,dx\;\longrightarrow\;\frac{a+b\,x+c\,\sqrt{1-(a+b\,x)^2}\,\text{Log}[f]}{b\,\left(1+c^2\,\text{Log}[f]^2\right)}\,f^{c\,\text{ArcSin}[a+b\,x]}$$

- **Program code:**

```
Int[f_^(c_.*ArcSin[a_.+b_.*x_]),x_Symbol] :=
  f^(c*ArcSin[a+b*x])*(a+b*x+c*Sqrt[1-(a+b*x)^2]*Log[f])/(b*(1+c^2*Log[f]^2)) /;
FreeQ[{a,b,c,f},x] && NonzeroQ[1+c^2*Log[f]^2]
```

$\int v \, \text{ArcSin}[u] \, dx$

- **Derivation: Integration by parts**
- **Rule: If u is free of inverse functions, then**

$$\int \text{ArcSin}[u] \, dx \longrightarrow x \, \text{ArcSin}[u] - \int \frac{x \, \partial_x u}{\sqrt{1 - u^2}} \, dx$$

- **Program code:**

```
Int[ArcSin[u_],x_Symbol] :=
  x*ArcSin[u] -
  Int[Regularize[x*D[u,x]/Sqrt[1-u^2],x],x] /;
InverseFunctionFreeQ[u,x] && Not[FunctionOfExponentialOfLinear[u,x]]
```

- **Derivation: Integration by parts**
- **Rule: If $m + 1 \neq 0 \land$ u is free of inverse functions, then**

$$\int x^m \, \text{ArcSin}[u] \, dx \longrightarrow \frac{x^{m+1} \, \text{ArcSin}[u]}{m+1} - \frac{1}{m+1} \int \frac{x^{m+1} \, \partial_x u}{\sqrt{1 - u^2}} \, dx$$

- **Program code:**

```
Int[x_^m_.*ArcSin[u_],x_Symbol] :=
  x^(m+1)*ArcSin[u]/(m+1) -
  Dist[1/(m+1),Int[Regularize[x^(m+1)*D[u,x]/Sqrt[1-u^2],x],x]] /;
FreeQ[m,x] && NonzeroQ[m+1] && InverseFunctionFreeQ[u,x] &&
    Not[FunctionOfQ[x^(m+1),u,x]] &&
    Not[FunctionOfExponentialOfLinear[u,x]]
```