$\int \text{ArcCos}[a+b\,x]^n\,dx$

- **Reference: G&R 2.814.1, CRC 442, A&S 4.4.59**
- **Derivation: Integration by parts**
- **Rule:**

$$\int \text{ArcCos}[a+b\,x]\,dx \longrightarrow \frac{(a+b\,x)\,\text{ArcCos}[a+b\,x]}{b}-\frac{\sqrt{1-(a+b\,x)^2}}{b}$$

- **Program code:**

```
Int[ArcCos[a_.+b_.*x_],x_Symbol] :=
  (a+b*x)*ArcCos[a+b*x]/b - Sqrt[1-(a+b*x)^2]/b /;
FreeQ[{a,b},x]
```

- **Reference: CRC 466**
- **Derivation: Iterated integration by parts**
- **Rule: If $n>1$, then**

$$\int \text{ArcCos}[a+b\,x]^n\,dx \longrightarrow \frac{(a+b\,x)\,\text{ArcCos}[a+b\,x]^n}{b}-\frac{n\sqrt{1-(a+b\,x)^2}\,\text{ArcCos}[a+b\,x]^{n-1}}{b}-n\,(n-1)\int \text{ArcCos}[a+b\,x]^{n-2}\,dx$$

- **Program code:**

```
Int[ArcCos[a_.+b_.*x_]^n_,x_Symbol] :=
  (a+b*x)*ArcCos[a+b*x]^n/b -
  n*Sqrt[1-(a+b*x)^2]*ArcCos[a+b*x]^(n-1)/b -
  Dist[n*(n-1),Int[ArcCos[a+b*x]^(n-2),x]] /;
FreeQ[{a,b},x] && RationalQ[n] && n>1
```

- **Derivation: Integration by substitution**
- **Basis:** $\frac{1}{\text{ArcCos}[z]} == -\frac{\text{Sin}[\text{ArcCos}[z]]}{\text{ArcCos}[z]}\,\text{ArcCos}'[z]$
- **Rule:**

$$\int \frac{1}{\text{ArcCos}[a+b\,x]}\,dx \longrightarrow -\frac{\text{SinIntegral}[\text{ArcCos}[a+b\,x]]}{b}$$

- **Program code:**

```
Int[1/ArcCos[a_.+b_.*x_],x_Symbol] :=
  -SinIntegral[ArcCos[a+b*x]]/b /;
FreeQ[{a,b},x]
```

- **Derivation: Integration by substitution**
- **Basis:** $\frac{1}{\text{ArcCos}[z]^2} = -\frac{\text{Sin}[\text{ArcCos}[z]]}{\text{ArcCos}[z]^2}\,\text{ArcCos}'[z]$
- **Rule:**

$$\int \frac{1}{\text{ArcCos}[a+b\,x]^2}\,dx \longrightarrow \frac{\sqrt{1-(a+b\,x)^2}}{b\,\text{ArcCos}[a+b\,x]} - \frac{\text{CosIntegral}[\text{ArcCos}[a+b\,x]]}{b}$$

- **Program code:**

```
Int[1/ArcCos[a_.+b_.*x_]^2,x_Symbol] :=
  Sqrt[1-(a+b*x)^2]/(b*ArcCos[a+b*x]) - CosIntegral[ArcCos[a+b*x]]/b /;
FreeQ[{a,b},x]
```

- **Derivation: Integration by substitution**
- **Basis:** $\frac{1}{\sqrt{\text{ArcCos}[z]}} = -\frac{\text{Sin}[\text{ArcCos}[z]]}{\sqrt{\text{ArcCos}[z]}}\,\text{ArcCos}'[z]$
- **Rule:**

$$\int \frac{1}{\sqrt{\text{ArcCos}[a+b\,x]}}\,dx \longrightarrow -\frac{1}{b}\sqrt{2\pi}\,\text{FresnelS}\left[\sqrt{\frac{2}{\pi}}\sqrt{\text{ArcCos}[a+b\,x]}\right]$$

- **Program code:**

```
Int[1/Sqrt[ArcCos[a_.+b_.*x_]],x_Symbol] :=
  -Sqrt[2*Pi]*FresnelS[Sqrt[2/Pi]*Sqrt[ArcCos[a+b*x]]]/b /;
FreeQ[{a,b},x]
```

- **Derivation: Integration by parts**
- **Rule:**

$$\int \sqrt{\text{ArcCos}[a+b\,x]}\,dx \longrightarrow \frac{(a+b\,x)\,\sqrt{\text{ArcCos}[a+b\,x]}}{b}-\frac{1}{b}\sqrt{\frac{\pi}{2}}\;\text{FresnelC}\Big[\sqrt{\frac{2}{\pi}}\,\sqrt{\text{ArcCos}[a+b\,x]}\Big]$$

- **Program code:**

```
Int[Sqrt[ArcCos[a_.+b_.*x_]],x_Symbol] :=
  (a+b*x)*Sqrt[ArcCos[a+b*x]]/b - Sqrt[Pi/2]*FresnelC[Sqrt[2/Pi]*Sqrt[ArcCos[a+b*x]]]/b /;
FreeQ[{a,b},x]
```

- **Derivation: Inverted iterated integration by parts**
- **Rule: If $n<-1 \land n\neq -2$, then**

$$\int \text{ArcCos}[a+b\,x]^n\,dx \longrightarrow \frac{(a+b\,x)\,\text{ArcCos}[a+b\,x]^{n+2}}{b\,(n+1)\,(n+2)}-\frac{\sqrt{1-(a+b\,x)^2}\,\text{ArcCos}[a+b\,x]^{n+1}}{b\,(n+1)}-\frac{1}{(n+1)\,(n+2)}\int \text{ArcCos}[a+b\,x]^{n+2}\,dx$$

- **Program code:**

```
Int[ArcCos[a_.+b_.*x_]^n_,x_Symbol] :=
  (a+b*x)*ArcCos[a+b*x]^(n+2)/(b*(n+1)*(n+2)) -
  Sqrt[1-(a+b*x)^2]*ArcCos[a+b*x]^(n+1)/(b*(n+1)) -
  Dist[1/((n+1)*(n+2)),Int[ArcCos[a+b*x]^(n+2),x]] /;
FreeQ[{a,b},x] && RationalQ[n] && n<-1 && n!=-2
```

- **Rule: If $n\notin\mathbb{Q} \lor -1<n<1$, then**

$$\int \text{ArcCos}[a+b\,x]^n\,dx \longrightarrow \frac{\text{ArcCos}[a+b\,x]^n}{2\,b}\left(\frac{\Gamma[n+1,\,i\,\text{ArcCos}[a+b\,x]]}{(i\,\text{ArcCos}[a+b\,x])^n}+\frac{\Gamma[n+1,\,-i\,\text{ArcCos}[a+b\,x]]}{(-i\,\text{ArcCos}[a+b\,x])^n}\right)$$

- **Program code:**

```
Int[ArcCos[a_.+b_.*x_]^n_,x_Symbol] :=
  ArcCos[a+b*x]^n/(2*b)*
    (Gamma[n+1,I*ArcCos[a+b*x]]/(I*ArcCos[a+b*x])^n +
     Gamma[n+1,-I*ArcCos[a+b*x]]/(-I*ArcCos[a+b*x])^n) /;
FreeQ[{a,b,n},x] && (Not[RationalQ[n]] || -1<n<1)
```

$$\int x^m \operatorname{ArcCos}[a+b\,x]\,dx$$

- **Reference: G&R 2.832, CRC 454, A&S 4.4.67**
- **Derivation: Integration by parts**
- **Rule: If $m+1 \neq 0$, then**

$$\int x^m \operatorname{ArcCos}[a+b\,x]\,dx \longrightarrow \frac{x^{m+1}\operatorname{ArcCos}[a+b\,x]}{m+1}+\frac{b}{m+1}\int\frac{x^{m+1}}{\sqrt{1-a^2-2\,a\,b\,x-b^2\,x^2}}\,dx$$

- **Program code:**

```
Int[x_^m_.*ArcCos[a_.+b_.*x_],x_Symbol] :=
  x^(m+1)*ArcCos[a+b*x]/(m+1) +
  Dist[b/(m+1),Int[x^(m+1)/Sqrt[1-a^2-2*a*b*x-b^2*x^2],x]] /;
FreeQ[{a,b,m},x] && NonzeroQ[m+1]
```

$$\int x^m \operatorname{ArcCos}[a\,x]^n \, dx$$

- **Rule:**

$$\int \frac{x}{\sqrt{\operatorname{ArcCos}[a\,x]}} \, dx \longrightarrow -\frac{\sqrt{\pi}}{2\,a^2} \operatorname{FresnelS}\left[\frac{2\sqrt{\operatorname{ArcCos}[a\,x]}}{\sqrt{\pi}}\right]$$

- **Program code:**

```
Int[x_/Sqrt[ArcCos[a_.*x_]],x_Symbol] :=
  -Sqrt[Pi]/(2*a^2)*FresnelS[2*Sqrt[ArcCos[a*x]]/Sqrt[Pi]] /;
FreeQ[a,x]
```

- **Rule:**

$$\int \frac{x}{\operatorname{ArcCos}[a\,x]^{3/2}} \, dx \longrightarrow \frac{2\,x\sqrt{1-a^2\,x^2}}{a\sqrt{\operatorname{ArcCos}[a\,x]}} - \frac{2\sqrt{\pi}}{a^2} \operatorname{FresnelC}\left[\frac{2\sqrt{\operatorname{ArcCos}[a\,x]}}{\sqrt{\pi}}\right]$$

- **Program code:**

```
Int[x_/ArcCos[a_.*x_]^(3/2),x_Symbol] :=
  2*x*Sqrt[1-a^2*x^2]/(a*Sqrt[ArcCos[a*x]]) - 2*Sqrt[Pi]/a^2*FresnelC[2*Sqrt[ArcCos[a*x]]/Sqrt[Pi]]
FreeQ[a,x]
```

- **Rule: If $n > 1$, then**

$$\int x \operatorname{ArcCos}[a\,x]^n \, dx \longrightarrow -\frac{n\,x\sqrt{1-a^2\,x^2}\;\operatorname{ArcCos}[a\,x]^{n-1}}{4\,a} - \frac{\operatorname{ArcCos}[a\,x]^n}{4\,a^2} + \frac{x^2 \operatorname{ArcCos}[a\,x]^n}{2} - \frac{n\,(n-1)}{4} \int x \operatorname{ArcCos}[a\,x]^{n-2} \, dx$$

- **Program code:**

```
Int[x_*ArcCos[a_.*x_]^n_,x_Symbol] :=
  -n*x*Sqrt[1-a^2*x^2]*ArcCos[a*x]^(n-1)/(4*a) -
  ArcCos[a*x]^n/(4*a^2) + x^2*ArcCos[a*x]^n/2 -
  Dist[n*(n-1)/4,Int[x*ArcCos[a*x]^(n-2),x]] /;
FreeQ[a,x] && RationalQ[n] && n>0
```

- **Rule: If $n < -1 \land n \neq -2$, then**

$$\int x \operatorname{ArcCos}[a x]^n \, dx \longrightarrow -\frac{x \sqrt{1-a^2 x^2} \operatorname{ArcCos}[a x]^{n+1}}{a (n+1)} - \frac{\operatorname{ArcCos}[a x]^{n+2}}{a^2 (n+1) (n+2)} + \frac{2 x^2 \operatorname{ArcCos}[a x]^{n+2}}{(n+1) (n+2)} - \frac{4}{(n+1) (n+2)} \int x \operatorname{ArcCos}[a x]^{n+2} \, dx$$

- **Program code:**

```
Int[x_*ArcCos[a_.*x_]^n_,x_Symbol] :=
  -x*Sqrt[1-a^2*x^2]*ArcCos[a*x]^(n+1)/(a*(n+1)) -
  ArcCos[a*x]^(n+2)/(a^2*(n+1)*(n+2)) +
  2*x^2*ArcCos[a*x]^(n+2)/((n+1)*(n+2)) -
  Dist[4/((n+1)*(n+2)),Int[x*ArcCos[a*x]^(n+2),x]] /;
FreeQ[a,x] && RationalQ[n] && n<-1 && n≠-2
```

- **Rule: If $n > 1$, then**

$$\int \frac{\operatorname{ArcCos}[a x]^n}{x^3} \, dx \longrightarrow \frac{a n \sqrt{1-a^2 x^2} \operatorname{ArcCos}[a x]^{n-1}}{2 x} - \frac{\operatorname{ArcCos}[a x]^n}{2 x^2} + \frac{a^2 n (n-1)}{2} \int \frac{\operatorname{ArcCos}[a x]^{n-2}}{x} \, dx$$

- **Program code:**

```
Int[ArcCos[a_.*x_]^n_/x_^3,x_Symbol] :=
  a*n*Sqrt[1-a^2*x^2]*ArcCos[a*x]^(n-1)/(2*x) -
  ArcCos[a*x]^n/(2*x^2) +
  Dist[a^2*n*(n-1)/2,Int[ArcCos[a*x]^(n-2)/x,x]] /;
FreeQ[a,x] && RationalQ[n] && n>1
```

- **Rule: If $m \in \mathbb{Z} \land m < -3 \land n > 1$, then**

$$\int x^m \text{ArcCos}[a\,x]^n\,dx \longrightarrow \frac{a\,n\,x^{m+2}\,\sqrt{1-a^2\,x^2}\,\text{ArcCos}[a\,x]^{n-1}}{(m+1)\,(m+2)}+\frac{x^{m+1}\,\text{ArcCos}[a\,x]^n}{(m+1)}-\frac{a^2\,(m+3)\,x^{m+3}\,\text{ArcCos}[a\,x]^n}{(m+1)\,(m+2)}+\frac{a^2\,(m+3)^2}{(m+1)\,(m+2)}\int x^{m+2}\,\text{ArcCos}[a\,x]^n\,dx+\frac{a^2\,n\,(n-1)}{(m+1)\,(m+2)}\int x^{m+2}\,\text{ArcCos}[a\,x]^{n-2}\,dx$$

- **Program code:**

```
Int[x_^m_*ArcCos[a_.*x_]^n_,x_Symbol] :=
  a*n*x^(m+2)*Sqrt[1-a^2*x^2]*ArcCos[a*x]^(n-1)/((m+1)*(m+2)) +
  x^(m+1)*ArcCos[a*x]^n/(m+1) -
  a^2*(m+3)*x^(m+3)*ArcCos[a*x]^n/((m+1)*(m+2)) +
  Dist[a^2*(m+3)^2/((m+1)*(m+2)),Int[x^(m+2)*ArcCos[a*x]^n,x]] +
  Dist[a^2*n*(n-1)/((m+1)*(m+2)),Int[x^(m+2)*ArcCos[a*x]^(n-2),x]] /;
FreeQ[a,x] && IntegerQ[m] && RationalQ[n] && m<-3 && n>1
```

- **Rule: If $m \in \mathbb{Z} \land m > 1 \land n < -1 \land n \neq -2$, then**

$$\int x^m \text{ArcCos}[a\,x]^n\,dx \longrightarrow -\frac{x^m\,\sqrt{1-a^2\,x^2}\,\text{ArcCos}[a\,x]^{n+1}}{a\,(n+1)}-\frac{m\,x^{m-1}\,\text{ArcCos}[a\,x]^{n+2}}{a^2\,(n+1)\,(n+2)}+\frac{(m+1)\,x^{m+1}\,\text{ArcCos}[a\,x]^{n+2}}{(n+1)\,(n+2)}-\frac{(m+1)^2}{(n+1)\,(n+2)}\int x^m\,\text{ArcCos}[a\,x]^{n+2}\,dx+\frac{m\,(m-1)}{a^2\,(n+1)\,(n+2)}\int x^{m-2}\,\text{ArcCos}[a\,x]^{n+2}\,dx$$

- **Program code:**

```
Int[x_^m_*ArcCos[a_.*x_]^n_,x_Symbol] :=
  -x^m*Sqrt[1-a^2*x^2]*ArcCos[a*x]^(n+1)/(a*(n+1)) -
  m*x^(m-1)*ArcCos[a*x]^(n+2)/(a^2*(n+1)*(n+2)) +
  (m+1)*x^(m+1)*ArcCos[a*x]^(n+2)/((n+1)*(n+2)) -
  Dist[(m+1)^2/((n+1)*(n+2)),Int[x^m*ArcCos[a*x]^(n+2),x]] +
  Dist[m*(m-1)/(a^2*(n+1)*(n+2)),Int[x^(m-2)*ArcCos[a*x]^(n+2),x]] /;
FreeQ[a,x] && IntegerQ[m] && RationalQ[n] && m>1 && n<-1 && n≠-2
```

- **Derivation: Integration by substitution**
- **Basis:** $\frac{\text{ArcCos}[a\,x^p]^n}{x} = -\frac{1}{p}\,\text{ArcCos}[a\,x^p]^n\,\text{Tan}[\text{ArcCos}[a\,x^p]]\,\partial_x\text{ArcCos}[a\,x^p]$
- **Rule: If** $n \in \mathbb{Z} \land n > 0$**, then**

$$\int \frac{\text{ArcCos}[a\,x^p]^n}{x}\,dx \longrightarrow -\frac{1}{p}\,\text{Subst}\Big[\int x^n\,\text{Tan}[x]\,dx,\,x,\,\text{ArcCos}[a\,x^p]\Big]$$

- **Program code:**

```
Int[ArcCos[a_.*x_^p_.]^n_./x_,x_Symbol] :=
  -Dist[1/p,Subst[Int[x^n*Tan[x],x],x,ArcCos[a*x^p]]] /;
FreeQ[{a,p},x] && IntegerQ[n] && n>0
```

- **Derivation: Integration by parts and substitution**
- **Basis: If** $m \in \mathbb{Z}$**,** $\frac{x^{m+1}\,\text{ArcCos}[a\,x]^{n-1}}{\sqrt{1-a^2\,x^2}} = -\frac{1}{a^{m+2}}\,\text{ArcCos}[a\,x]^{n-1}\,\text{Cos}[\text{ArcCos}[a\,x]]^{m+1}\,\partial_x\text{ArcCos}[a\,x]$
- **Rule: If** $m \in \mathbb{Z} \land m \neq -1$**, then**

$$\int x^m\,\text{ArcCos}[a\,x]^n\,dx \longrightarrow \frac{x^{m+1}\,\text{ArcCos}[a\,x]^n}{m+1} - \frac{n}{a^{m+1}\,(m+1)}\,\text{Subst}\Big[\int x^{n-1}\,\text{Cos}[x]^{m+1}\,dx,\,x,\,\text{ArcCos}[a\,x]\Big]$$

- **Program code:**

```
Int[x_^m_.*ArcCos[a_.*x_]^n_,x_Symbol] :=
  x^(m+1)*ArcCos[a*x]^n/(m+1) -
  Dist[n/(a^(m+1)*(m+1)),Subst[Int[x^(n-1)*Cos[x]^(m+1),x],x,ArcCos[a*x]]] /;
FreeQ[{a,n},x] && IntegerQ[m] && m≠-1
```

$$\int (a + b\,\text{ArcCos}[c + d\,x])^n\,dx$$

- **Derivation: Integration by substitution**

- **Basis:** $(a + b\,\text{ArcCos}[c + d\,x])^n = -\frac{1}{d}\,(a + b\,\text{ArcCos}[c + d\,x])^n\,\text{Sin}[\text{ArcCos}[c + d\,x]]\,\partial_x \text{ArcCos}[c + d\,x]$

- **Rule: If** $n \notin \mathbb{Z}$**, then**

$$\int (a + b\,\text{ArcCos}[c + d\,x])^n\,dx \longrightarrow -\frac{1}{d}\,\text{Subst}\left[\int (a + b\,x)^n\,\text{Sin}[x]\,dx,\ x,\ \text{ArcCos}[c + d\,x]\right]$$

- **Program code:**

```
Int[(a_+b_.*ArcCos[c_.+d_.*x_])^n_,x_Symbol] :=
  -Dist[1/d,Subst[Int[(a+b*x)^n*Sin[x],x],x,ArcCos[c+d*x]]] /;
FreeQ[{a,b,c,d},x] && Not[IntegerQ[n]]
```

$$\int x^m\,(a+b\,\text{ArcCos}[c+d\,x])^n\,dx$$

- **Derivation: Integration by substitution**
- **Basis: If** $m\in\mathbb{Z}$, $x^m\,(a+b\,\text{ArcCos}[c+d\,x])^n =$
$$-\frac{1}{d^{m+1}}\,(a+b\,\text{ArcCos}[c+d\,x])^n\,(\text{Cos}[\text{ArcCos}[c+d\,x]]-c)^m\,\text{Sin}[\text{ArcCos}[c+d\,x]]\,\partial_x\text{ArcCos}[c+d\,x]$$
- **Rule: If** $m\in\mathbb{Z}\land n\notin\mathbb{Z}\land m>0$, **then**

$$\int x^m\,(a+b\,\text{ArcCos}[c+d\,x])^n\,dx\;\longrightarrow\;-\frac{1}{d^{m+1}}\,\text{Subst}\Big[\int (a+b\,x)^n\,(\text{Cos}[x]-c)^m\,\text{Sin}[x]\,dx,\,x,\,\text{ArcCos}[c+d\,x]\Big]$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*ArcCos[c_.+d_.*x_])^n_,x_Symbol] :=
  -Dist[1/d^(m+1),Subst[Int[(a+b*x)^n*(Cos[x]-c)^m*Sin[x],x],x,ArcCos[c+d*x]]] /;
FreeQ[{a,b,c,d},x] && IntegerQ[m] && Not[IntegerQ[n]] && m>0
```

$$\int \frac{x \operatorname{ArcCos}[a+b x]^n}{\sqrt{1-(a+b x)^2}} \, dx$$

- **Derivation: Integration by parts**
- **Rule: If $n > 1$, then**

$$\int \frac{x \operatorname{ArcCos}[a+b x]^n}{\sqrt{1-(a+b x)^2}} \, dx \longrightarrow$$

$$-\frac{\sqrt{1-(a+b x)^2} \operatorname{ArcCos}[a+b x]^n}{b^2}-\frac{n}{b} \int \operatorname{ArcCos}[a+b x]^{n-1} \, dx-\frac{a}{b} \int \frac{\operatorname{ArcCos}[a+b x]^n}{\sqrt{1-(a+b x)^2}} \, dx$$

- **Program code:**

```
Int[x_*ArcCos[a_.+b_.*x_]^n_/Sqrt[u_],x_Symbol] :=
  -Sqrt[u]*ArcCos[a+b*x]^n/b^2 -
  Dist[n/b,Int[ArcCos[a+b*x]^(n-1),x]] -
  Dist[a/b,Int[ArcCos[a+b*x]^n/Sqrt[u],x]] /;
FreeQ[{a,b},x] && ZeroQ[u-1+(a+b*x)^2] && RationalQ[n] && n>1
```

$$\int u \operatorname{ArcCos}\left[\frac{c}{a+b\,x^n}\right]^m dx$$

- **Derivation: Algebraic simplification**

- **Basis:** $\operatorname{ArcCos}[z] = \operatorname{ArcSec}\left[\frac{1}{z}\right]$

- **Rule:**

$$\int u \operatorname{ArcCos}\left[\frac{c}{a+b\,x^n}\right]^m dx \longrightarrow \int u \operatorname{ArcSec}\left[\frac{a}{c}+\frac{b\,x^n}{c}\right]^m dx$$

- **Program code:**

```
Int[u_.*ArcCos[c_./(a_.+b_.*x_^n_.)]^m_.,x_Symbol] :=
  Int[u*ArcSec[a/c+b*x^n/c]^m,x] /;
FreeQ[{a,b,c,n,m},x]
```

$$\int f^{c \text{ ArcCos}[a+b x]} \, dx$$

- **Rule: If $1 + c^2 \text{ Log}[f]^2 \neq 0$, then**

$$\int f^{c \text{ ArcCos}[a+b x]} \, dx \longrightarrow \frac{a + b x - c \sqrt{1 - (a + b x)^2} \text{ Log}[f]}{b \left(1 + c^2 \text{ Log}[f]^2\right)} f^{c \text{ ArcCos}[a+b x]}$$

- **Program code:**

```
Int[f_^(c_.*ArcCos[a_.+b_.*x_]),x_Symbol] :=
  f^(c*ArcCos[a+b*x])*(a+b*x-c*Sqrt[1-(a+b*x)^2]*Log[f])/(b*(1+c^2*Log[f]^2)) /;
FreeQ[{a,b,c,f},x] && NonzeroQ[1+c^2*Log[f]^2]
```

$\int v \, \text{ArcCos}[u] \, dx$

- **Derivation: Integration by parts**
- **Rule: If u is free of inverse functions, then**

$$\int \text{ArcCos}[u] \, dx \longrightarrow x \, \text{ArcCos}[u] + \int \frac{x \, \partial_x u}{\sqrt{1-u^2}} \, dx$$

- **Program code:**

```
Int[ArcCos[u_],x_Symbol] :=
  x*ArcCos[u] +
  Int[Regularize[x*D[u,x]/Sqrt[1-u^2],x],x] /;
InverseFunctionFreeQ[u,x] && Not[FunctionOfExponentialOfLinear[u,x]]
```

- **Derivation: Integration by parts**
- **Rule: If $m+1 \neq 0 \land$ u is free of inverse functions, then**

$$\int x^m \, \text{ArcCos}[u] \, dx \longrightarrow \frac{x^{m+1} \, \text{ArcCos}[u]}{m+1} + \frac{1}{m+1} \int \frac{x^{m+1} \, \partial_x u}{\sqrt{1-u^2}} \, dx$$

- **Program code:**

```
Int[x_^m_.*ArcCos[u_],x_Symbol] :=
  x^(m+1)*ArcCos[u]/(m+1) +
  Dist[1/(m+1),Int[Regularize[x^(m+1)*D[u,x]/Sqrt[1-u^2],x],x]] /;
FreeQ[m,x] && NonzeroQ[m+1] && InverseFunctionFreeQ[u,x] &&
    Not[FunctionOfQ[x^(m+1),u,x]] &&
    Not[FunctionOfExponentialOfLinear[u,x]]
```