$\int \left(f^{a+b x}\right)^p \text{Sinh}[c+d x]^n \, dx$

- **Reference: CRC 533h**

- **Rule: If $d^2 - b^2 p^2 \text{Log}[f]^2 \neq 0$, then**

$$\int \left(f^{a+b x}\right)^p \text{Sinh}[c+d x] \, dx \longrightarrow$$
$$-\frac{b\, p\, \text{Log}[f] \left(f^{a+b x}\right)^p \text{Sinh}[c+d x]}{d^2 - b^2 p^2 \text{Log}[f]^2} + \frac{d \left(f^{a+b x}\right)^p \text{Cosh}[c+d x]}{d^2 - b^2 p^2 \text{Log}[f]^2}$$

- **Program code:**

```
Int[(f_^(a_.+b_.*x_))^p_.*Sinh[c_.+d_.*x_],x_Symbol] :=
  -b*p*Log[f]*(f^(a+b*x))^p*Sinh[c+d*x]/(d^2-b^2*p^2*Log[f]^2) +
  d*(f^(a+b*x))^p*Cosh[c+d*x]/(d^2-b^2*p^2*Log[f]^2) /;
FreeQ[{a,b,c,d,f,p},x] && NonzeroQ[d^2-b^2*p^2*Log[f]^2]
```

- **Reference: CRC 538h**

```
Int[(f_^(a_.+b_.*x_))^p_.*Cosh[c_.+d_.*x_],x_Symbol] :=
  -b*p*Log[f]*(f^(a+b*x))^p*Cosh[c+d*x]/(d^2-b^2*p^2*Log[f]^2) +
  d*(f^(a+b*x))^p*Sinh[c+d*x]/(d^2-b^2*p^2*Log[f]^2) /;
FreeQ[{a,b,c,d,f,p},x] && NonzeroQ[d^2-b^2*p^2*Log[f]^2]
```

- **Reference: CRC 542h**

- **Rule: If $d^2 n^2 - b^2 p^2 \text{Log}[f]^2 \neq 0 \land n > 1$, then**

$$\int \left(f^{a+b x}\right)^p \text{Sinh}[c+d x]^n \, dx \longrightarrow -\frac{b\, p\, \text{Log}[f] \left(f^{a+b x}\right)^p \text{Sinh}[c+d x]^n}{d^2 n^2 - b^2 p^2 \text{Log}[f]^2} +$$
$$\frac{d\, n \left(f^{a+b x}\right)^p \text{Cosh}[c+d x]\, \text{Sinh}[c+d x]^{n-1}}{d^2 n^2 - b^2 p^2 \text{Log}[f]^2} - \frac{n\,(n-1)\, d^2}{d^2 n^2 - b^2 p^2 \text{Log}[f]^2} \int \left(f^{a+b x}\right)^p \text{Sinh}[c+d x]^{n-2} \, dx$$

- **Program code:**

```
Int[(f_^(a_.+b_.*x_))^p_.*Sinh[c_.+d_.*x_]^n_,x_Symbol] :=
  -b*p*Log[f]*(f^(a+b*x))^p*Sinh[c+d*x]^n/(d^2*n^2-b^2*p^2*Log[f]^2) +
  d*n*(f^(a+b*x))^p*Cosh[c+d*x]*Sinh[c+d*x]^(n-1)/(d^2*n^2-b^2*p^2*Log[f]^2) -
  Dist[n*(n-1)*d^2/(d^2*n^2-b^2*p^2*Log[f]^2),Int[(f^(a+b*x))^p*Sinh[c+d*x]^(n-2),x]] /;
FreeQ[{a,b,c,d,f,p},x] && NonzeroQ[d^2*n^2-b^2*p^2*Log[f]^2] && RationalQ[n] && n>1
```

- **Reference: CRC 543h**

```
Int[(f_^(a_.+b_.*x_))^p_.*Cosh[c_.+d_.*x_]^n_,x_Symbol] :=
  -b*p*Log[f]*(f^(a+b*x))^p*Cosh[c+d*x]^n/(d^2*n^2-b^2*p^2*Log[f]^2) +
  d*n*(f^(a+b*x))^p*Sinh[c+d*x]*Cosh[c+d*x]^(n-1)/(d^2*n^2-b^2*p^2*Log[f]^2) +
  Dist[n*(n-1)*d^2/(d^2*n^2-b^2*p^2*Log[f]^2),Int[(f^(a+b*x))^p*Cosh[c+d*x]^(n-2),x]] /;
FreeQ[{a,b,c,d,f,p},x] && NonzeroQ[d^2*n^2-b^2*p^2*Log[f]^2] && RationalQ[n] && n>1
```

- **Reference: CRC 551h when $d^2 (n+2)^2 - b^2 p^2 \text{Log}[f]^2 = 0$**

- **Rule: If $d^2 (n+2)^2 - b^2 p^2 \text{Log}[f]^2 = 0 \wedge n+1 \neq 0 \wedge n+2 \neq 0$, then**

$$\int \left(f^{a+b x}\right)^p \text{Sinh}[c+d x]^n \, dx \longrightarrow$$
$$-\frac{b\, p\, \text{Log}[f] \left(f^{a+b x}\right)^p \text{Sinh}[c+d x]^{n+2}}{d^2 (n+1) (n+2)} + \frac{\left(f^{a+b x}\right)^p \text{Cosh}[c+d x]\, \text{Sinh}[c+d x]^{n+1}}{d (n+1)}$$

- **Program code:**

```
Int[(f_^(a_.+b_.*x_))^p_.*Sinh[c_.+d_.*x_]^n_,x_Symbol] :=
  -b*p*Log[f]*(f^(a+b*x))^p*Sinh[c+d*x]^(n+2)/(d^2*(n+1)*(n+2)) +
  (f^(a+b*x))^p*Cosh[c+d*x]*Sinh[c+d*x]^(n+1)/(d*(n+1)) /;
FreeQ[{a,b,c,d,f,n,p},x] && ZeroQ[d^2*(n+2)^2-b^2*p^2*Log[f]^2] && NonzeroQ[n+1] && NonzeroQ[n+2]
```

- **Reference: CRC 552h when $d^2 (n+2)^2 - b^2 p^2 \text{Log}[f]^2 = 0$**

```
Int[(f_^(a_.+b_.*x_))^p_.*Cosh[c_.+d_.*x_]^n_,x_Symbol] :=
  b*p*Log[f]*(f^(a+b*x))^p*Cosh[c+d*x]^(n+2)/(d^2*(n+1)*(n+2)) -
  (f^(a+b*x))^p*Sinh[c+d*x]*Cosh[c+d*x]^(n+1)/(d*(n+1)) /;
FreeQ[{a,b,c,d,f,n,p},x] && ZeroQ[d^2*(n+2)^2-b^2*p^2*Log[f]^2] && NonzeroQ[n+1] && NonzeroQ[n+2]
```

- **Reference: CRC 551h, CRC 542h inverted**

- **Rule: If $d^2 (n+2)^2 - b^2 p^2 \text{Log}[f]^2 \neq 0 \wedge n < -1 \wedge n \neq -2$, then**

$$\int \left(f^{a+b x}\right)^p \text{Sinh}[c+d x]^n \, dx \longrightarrow -\frac{b\, p\, \text{Log}[f] \left(f^{a+b x}\right)^p \text{Sinh}[c+d x]^{n+2}}{d^2 (n+1) (n+2)} +$$
$$\frac{\left(f^{a+b x}\right)^p \text{Cosh}[c+d x]\, \text{Sinh}[c+d x]^{n+1}}{d (n+1)} - \frac{d^2 (n+2)^2 - b^2 p^2 \text{Log}[f]^2}{d^2 (n+1) (n+2)} \int \left(f^{a+b x}\right)^p \text{Sinh}[c+d x]^{n+2} \, dx$$

- **Program code:**

```
Int[(f_^(a_.+b_.*x_))^p_.*Sinh[c_.+d_.*x_]^n_,x_Symbol] :=
  -b*p*Log[f]*(f^(a+b*x))^p*Sinh[c+d*x]^(n+2)/(d^2*(n+1)*(n+2)) +
  (f^(a+b*x))^p*Cosh[c+d*x]*Sinh[c+d*x]^(n+1)/(d*(n+1)) -
  Dist[(d^2*(n+2)^2-b^2*p^2*Log[f]^2)/(d^2*(n+1)*(n+2)),Int[(f^(a+b*x))^p*Sinh[c+d*x]^(n+2),x]] /;
FreeQ[{a,b,c,d,f,p},x] && NonzeroQ[d^2*(n+2)^2-b^2*p^2*Log[f]^2] && RationalQ[n] && n<-1 && n≠-2
```

Reference: CRC 552h, CRC 543h inverted

```
Int[(f_^(a_.+b_.*x_))^p_.*Cosh[c_.+d_.*x_]^n_,x_Symbol] :=
  b*p*Log[f]*(f^(a+b*x))^p*Cosh[c+d*x]^(n+2)/(d^2*(n+1)*(n+2)) -
  (f^(a+b*x))^p*Sinh[c+d*x]*Cosh[c+d*x]^(n+1)/(d*(n+1)) +
  Dist[(d^2*(n+2)^2-b^2*p^2*Log[f]^2)/(d^2*(n+1)*(n+2)),Int[(f^(a+b*x))^p*Cosh[c+d*x]^(n+2),x]] /;
FreeQ[{a,b,c,d,f,p},x] && NonzeroQ[d^2*(n+2)^2-b^2*p^2*Log[f]^2] && RationalQ[n] && n<-1 && n≠-2
```

$$\int \left(f^{a+b\,x}\right)^p \,\text{Sech}[c+d\,x]^n \, dx$$

- **Reference:** CRC 552h with $b^2 p^2 \,\text{Log}[f]^2 - d^2 (n-2)^2 = 0$

- **Rule:** If $b^2 p^2 \,\text{Log}[f]^2 - d^2 (n-2)^2 = 0 \land n-1 \neq 0 \land n-2 \neq 0$, then

$$\int \left(f^{a+b\,x}\right)^p \,\text{Sech}[c+d\,x]^n \, dx \longrightarrow$$
$$\frac{b\,p\,\text{Log}[f] \left(f^{a+b\,x}\right)^p \text{Sech}[c+d\,x]^{n-2}}{d^2 (n-1)(n-2)} + \frac{\left(f^{a+b\,x}\right)^p \text{Sech}[c+d\,x]^{n-1} \,\text{Sinh}[c+d\,x]}{d\,(n-1)}$$

- **Program code:**

```
Int[(f_^(a_.+b_.*x_))^p_.*Sech[c_.+d_.*x_]^n_,x_Symbol] :=
  b*p*Log[f]*(f^(a+b*x))^p*Sech[c+d*x]^(n-2)/(d^2*(n-1)*(n-2)) +
  (f^(a+b*x))^p*Sech[c+d*x]^(n-1)*Sinh[c+d*x]/(d*(n-1)) /;
FreeQ[{a,b,c,d,f,p,n},x] && ZeroQ[b^2*p^2*Log[f]^2-d^2*(n-2)^2] && NonzeroQ[n-1] && NonzeroQ[n-2]
```

- **Reference:** CRC 551h with $b^2 p^2 \,\text{Log}[f]^2 - d^2 (n-2)^2 = 0$

```
Int[(f_^(a_.+b_.*x_))^p_.*Csch[c_.+d_.*x_]^n_,x_Symbol] :=
  -b*p*Log[f]*(f^(a+b*x))^p*Csch[c+d*x]^(n-2)/(d^2*(n-1)*(n-2)) -
  (f^(a+b*x))^p*Csch[c+d*x]^(n-1)*Cosh[c+d*x]/(d*(n-1)) /;
FreeQ[{a,b,c,d,f,p,n},x] && ZeroQ[b^2*p^2*Log[f]^2-d^2*(n-2)^2] && NonzeroQ[n-1] && NonzeroQ[n-2]
```

- **Reference:** CRC 552h

- **Rule:** If $b^2 p^2 \,\text{Log}[f]^2 - d^2 (n-2)^2 \neq 0 \land n > 1 \land n \neq 2$, then

$$\int \left(f^{a+b\,x}\right)^p \,\text{Sech}[c+d\,x]^n \, dx \longrightarrow \frac{b\,p\,\text{Log}[f] \left(f^{a+b\,x}\right)^p \text{Sech}[c+d\,x]^{n-2}}{d^2 (n-1)(n-2)} +$$
$$\frac{\left(f^{a+b\,x}\right)^p \text{Sech}[c+d\,x]^{n-1} \,\text{Sinh}[c+d\,x]}{d\,(n-1)} - \frac{b^2 p^2 \,\text{Log}[f]^2 - d^2 (n-2)^2}{d^2 (n-1)(n-2)} \int \left(f^{a+b\,x}\right)^p \text{Sech}[c+d\,x]^{n-2} \, dx$$

- **Program code:**

```
Int[(f_^(a_.+b_.*x_))^p_.*Sech[c_.+d_.*x_]^n_,x_Symbol] :=
  b*p*Log[f]*(f^(a+b*x))^p*Sech[c+d*x]^(n-2)/(d^2*(n-1)*(n-2)) +
  (f^(a+b*x))^p*Sech[c+d*x]^(n-1)*Sinh[c+d*x]/(d*(n-1)) -
  Dist[(b^2*p^2*Log[f]^2-d^2*(n-2)^2)/(d^2*(n-1)*(n-2)),
    Int[(f^(a+b*x))^p*Sech[c+d*x]^(n-2),x]] /;
FreeQ[{a,b,c,d,f,p},x] && NonzeroQ[b^2*p^2*Log[f]^2-d^2*(n-2)^2] &&
  RationalQ[n] && n>1 && n≠2
```

- **Reference: CRC 551h**

```
Int[(f_^(a_.+b_.*x_))^p_.*Csch[c_.+d_.*x_]^n_,x_Symbol] :=
  -b*p*Log[f]*(f^(a+b*x))^p*Csch[c+d*x]^(n-2)/(d^2*(n-1)*(n-2)) -
  (f^(a+b*x))^p*Csch[c+d*x]^(n-1)*Cosh[c+d*x]/(d*(n-1)) +
  Dist[(b^2*p^2*Log[f]^2-d^2*(n-2)^2)/(d^2*(n-1)*(n-2)),
    Int[(f^(a+b*x))^p*Csch[c+d*x]^(n-2),x]] /;
FreeQ[{a,b,c,d,f,p},x] && NonzeroQ[b^2*p^2*Log[f]^2-d^2*(n-2)^2] &&
  RationalQ[n] && n>1 && n≠2
```

$$\int x^m \left(f^{a+b x}\right)^p \operatorname{Sinh}[c+d x]^n \, dx$$

- **Derivation: Integration by parts**
- **Note: Each term of the sum $x^{m-1}\, u$ will be similar in form to the original integrand, but the degree of the monomial will be smaller by one.**
- **Rule: If $m>0 \land n\in\mathbb{Z} \land n>0$, let $u=\int (f^{a+b x})^p \operatorname{Sinh}[c+d x]\, dx$, then**

$$\int x^m \left(f^{a+b x}\right)^p \operatorname{Sinh}[c+d x]^n \, dx \longrightarrow x^m\, u - m \int x^{m-1}\, u \, dx$$

- **Program code:**

```
Int[x_^m_.*(f_^(a_.+b_.*x_))^p_.*Sinh[c_.+d_.*x_]^n_.,x_Symbol] :=
  Module[{u=Block[{ShowSteps=False,StepCounter=Null}, Int[(f^(a+b*x))^p*Sinh[c+d*x]^n,x]]},
  x^m*u - Dist[m,Int[x^(m-1)*u,x]]] /;
FreeQ[{a,b,c,d,f,p},x] && RationalQ[m] && m>0 && IntegerQ[n] && n>0
```

```
Int[x_^m_.*(f_^(a_.+b_.*x_))^p_.*Cosh[c_.+d_.*x_]^n_.,x_Symbol] :=
  Module[{u=Block[{ShowSteps=False,StepCounter=Null}, Int[(f^(a+b*x))^p*Cosh[c+d*x]^n,x]]},
  x^m*u - Dist[m,Int[x^(m-1)*u,x]]] /;
FreeQ[{a,b,c,d,f,p},x] && RationalQ[m] && m>0 && IntegerQ[n] && n>0
```

$$\int f^v \operatorname{Sinh}[w]^n \, dx$$

- **Derivation: Algebraic expansion**
- **Basis:** $\operatorname{Sinh}[z] = \frac{e^z}{2} - \frac{1}{2 e^z}$
- **Rule: If v and w are quadratic polynomials in x, then**

$$\int f^v \operatorname{Sinh}[w] \, dx \longrightarrow \frac{1}{2} \int f^v e^w \, dx - \frac{1}{2} \int \frac{f^v}{e^w} \, dx$$

- **Program code:**

```
Int[f_^v_*Sinh[w_],x_Symbol] :=
  Dist[1/2,Int[f^v*E^w,x]] -
  Dist[1/2,Int[f^v/E^w,x]] /;
FreeQ[f,x] && PolynomialQ[v,x] && Exponent[v,x]≤2 && PolynomialQ[w,x] && Exponent[w,x]≤2
```

- **Basis:** $\operatorname{Cosh}[z] = \frac{e^z}{2} + \frac{1}{2 e^z}$

```
Int[f_^v_*Cosh[w_],x_Symbol] :=
  Dist[1/2,Int[f^v*E^w,x]] +
  Dist[1/2,Int[f^v/E^w,x]] /;
FreeQ[f,x] && PolynomialQ[v,x] && Exponent[v,x]≤2 && PolynomialQ[w,x] && Exponent[w,x]≤2
```

- **Derivation: Algebraic expansion**
- **Basis:** $\operatorname{Sinh}[z] = \frac{1}{2} \left(e^z - \frac{1}{e^z}\right)$
- **Rule: If** $n \in \mathbb{Z} \land n > 0 \land$ **v and w are quadratic polynomials in x, then**

$$\int f^v \operatorname{Sinh}[w]^n \, dx \longrightarrow \frac{1}{2^n} \int f^v \left(e^w - \frac{1}{e^w}\right)^n dx$$

- **Program code:**

```
Int[f_^v_*Sinh[w_]^n_,x_Symbol] :=
  Dist[1/2^n,Int[f^v*(E^w-1/E^w)^n,x]] /;
FreeQ[f,x] && IntegerQ[n] && n>0 && PolynomialQ[v,x] && Exponent[v,x]≤2 &&
  PolynomialQ[w,x] && Exponent[w,x]≤2
```

- **Basis:** $\operatorname{Cosh}[z] = \frac{1}{2} \left(e^z + \frac{1}{e^z}\right)$

```
Int[f_^v_*Cosh[w_]^n_,x_Symbol] :=
  Dist[1/2^n,Int[f^v*(E^w+1/E^w)^n,x]] /;
FreeQ[f,x] && IntegerQ[n] && n>0 && PolynomialQ[v,x] && Exponent[v,x]≤2 &&
  PolynomialQ[w,x] && Exponent[w,x]≤2
```