$\int \text{PolyLog}[n, a\,(b\,x^p)^q]\,dx$

- **Derivation: Integration by parts**
- **Rule: If $n > 0$, then**

$$\int \text{PolyLog}[n, a\,(b\,x^p)^q]\,dx \longrightarrow x\,\text{PolyLog}[n, a\,(b\,x^p)^q] - p\,q\int \text{PolyLog}[n-1, a\,(b\,x^p)^q]\,dx$$

- **Program code:**

```
Int[PolyLog[n_,a_.*(b_.*x_^p_.)^q_.],x_Symbol] :=
  x*PolyLog[n,a*(b*x^p)^q] -
  Dist[p*q,Int[PolyLog[n-1,a*(b*x^p)^q],x]] /;
FreeQ[{a,b,p,q},x] && RationalQ[n] && n>0
```

- **Derivation: Inverted integration by parts**
- **Rule: If $n < -1$, then**

$$\int \text{PolyLog}[n, a\,(b\,x^p)^q]\,dx \longrightarrow \frac{x\,\text{PolyLog}[n+1, a\,(b\,x^p)^q]}{p\,q} - \frac{1}{p\,q}\int \text{PolyLog}[n+1, a\,(b\,x^p)^q]\,dx$$

- **Program code:**

```
Int[PolyLog[n_,a_.*(b_.*x_^p_.)^q_.],x_Symbol] :=
  x*PolyLog[n+1,a*(b*x^p)^q]/(p*q) -
  Dist[1/(p*q),Int[PolyLog[n+1,a*(b*x^p)^q],x]] /;
FreeQ[{a,b,p,q},x] && RationalQ[n] && n<-1
```

$\int x^m \text{PolyLog}[n, a (b x^p)^q] \, dx$

- **Derivation: Primitive rule**
- **Basis:** $\frac{\partial \text{Li}_n(z)}{\partial z} = \frac{\text{Li}_{n-1}(z)}{z}$
- **Rule:**

$$\int \frac{\text{PolyLog}[n, a (b x^p)^q]}{x} \, dx \longrightarrow \frac{\text{PolyLog}[n+1, a (b x^p)^q]}{p q}$$

- **Program code:**

```
Int[PolyLog[n_,a_.*(b_.*x_^p_.)^q_.]/x_,x_Symbol] :=
  PolyLog[n+1,a*(b*x^p)^q]/(p*q) /;
FreeQ[{a,b,n,p,q},x]
```

- **Derivation: Integration by parts**
- **Rule: If** $n > 0 \land m + 1 \neq 0$**, then**

$$\int x^m \text{PolyLog}[n, a (b x^p)^q] \, dx \longrightarrow \frac{x^{m+1} \text{PolyLog}[n, a (b x^p)^q]}{m+1} - \frac{p q}{m+1} \int x^m \text{PolyLog}[n-1, a (b x^p)^q] \, dx$$

- **Program code:**

```
Int[x_^m_.*PolyLog[n_,a_.*(b_.*x_^p_.)^q_.],x_Symbol] :=
  x^(m+1)*PolyLog[n,a*(b*x^p)^q]/(m+1) -
  Dist[p*q/(m+1),Int[x^m*PolyLog[n-1,a*(b*x^p)^q],x]] /;
FreeQ[{a,b,m,p,q},x] && RationalQ[n] && n>0 && NonzeroQ[m+1]
```

- **Derivation: Inverted integration by parts**
- **Rule: If** $n < -1 \land m + 1 \neq 0$**, then**

$$\int x^m \text{PolyLog}[n, a (b x^p)^q] \, dx \longrightarrow \frac{x^{m+1} \text{PolyLog}[n+1, a (b x^p)^q]}{p q} - \frac{m+1}{p q} \int x^m \text{PolyLog}[n+1, a (b x^p)^q] \, dx$$

- **Program code:**

```
Int[x_^m_.*PolyLog[n_,a_.*(b_.*x_^p_.)^q_.],x_Symbol] :=
  x^(m+1)*PolyLog[n+1,a*(b*x^p)^q]/(p*q) -
  Dist[(m+1)/(p*q),Int[x^m*PolyLog[n+1,a*(b*x^p)^q],x]] /;
FreeQ[{a,b,m,p,q},x] && RationalQ[n] && n<-1 && NonzeroQ[m+1]
```

- **Derivation: Integration by substitution**
- **Rule:**

$$\int \frac{\text{PolyLog}[n,\,u]}{a+b\,x}\,dx \longrightarrow \frac{1}{b}\,\text{Subst}\left[\int \frac{\text{PolyLog}\left[n,\,\text{Regularize}\left[\text{Subst}\left[u,\,x,\,-\frac{a}{b}+\frac{x}{b}\right],\,x\right]\right]}{x}\,dx,\,x,\,a+b\,x\right]$$

- **Program code:**

```
Int[PolyLog[n_,u_]/(a_+b_.*x_),x_Symbol] :=
  Dist[1/b,Subst[Int[PolyLog[n,Regularize[Subst[u,x,-a/b+x/b],x]]/x,x],x,a+b*x]] /;
FreeQ[{a,b,n},x]
```

$$\int \text{PolyLog}[n, c\,(a + b\,x)^p]\,dx$$

- **Derivation: Integration by parts**
- **Rule: If $n > 0$, then**

$$\int \text{PolyLog}[n, c\,(a + b\,x)^p]\,dx \longrightarrow x\,\text{PolyLog}[n, c\,(a + b\,x)^p] - p\int \text{PolyLog}[n-1, c\,(a + b\,x)^p]\,dx + a\,p\int \frac{\text{PolyLog}[n-1, c\,(a + b\,x)^p]}{a + b\,x}\,dx$$

- **Program code:**

```
Int[PolyLog[n_,c_.*(a_.+b_.*x_)^p_.],x_Symbol] :=
  x*PolyLog[n,c*(a+b*x)^p] -
  Dist[p,Int[PolyLog[n-1,c*(a+b*x)^p],x]] +
  Dist[a*p,Int[PolyLog[n-1,c*(a+b*x)^p]/(a+b*x),x]] /;
FreeQ[{a,b,c,p},x] && RationalQ[n] && n>0
```

$$\int x^m \,\mathrm{PolyLog}[n,\, c\,(a+b\,x)^p]\,dx$$

- **Derivation: Integration by parts**
- **Rule: If** $n>0 \land m\in\mathbb{Z} \land m>0$, **then**

$$\int x^m \,\mathrm{PolyLog}[n,\, c\,(a+b\,x)^p]\,dx \longrightarrow$$
$$\frac{x^{m+1}\,\mathrm{PolyLog}[n,\, c\,(a+b\,x)^p]}{m+1} - \frac{b\,p}{m+1}\int \frac{x^{m+1}\,\mathrm{PolyLog}[n-1,\, c\,(a+b\,x)^p]}{a+b\,x}\,dx$$

- **Program code:**

```
Int[x_^m_.*PolyLog[n_,c_.*(a_.+b_.*x_)^p_.],x_Symbol] :=
  x^(m+1)*PolyLog[n,c*(a+b*x)^p]/(m+1) -
  Dist[b*p/(m+1),Int[x^(m+1)*PolyLog[n-1,c*(a+b*x)^p]/(a+b*x),x]] /;
FreeQ[{a,b,c,m,p},x] && RationalQ[n] && n>0 && IntegerQ[m] && m>0
```

$$\int \text{PolyLog}\left[n, c\, f^{a+b\, x}\right] \, dx$$

- **Derivation: Primitive rule**
- **Basis:** $\partial_z \text{PolyLog}[n, z] = \frac{\text{PolyLog}[n-1, z]}{z}$
- **Rule:**

$$\int \text{PolyLog}\left[n, c\, f^{a+b\, x}\right] \, dx \longrightarrow \frac{\text{PolyLog}\left[n+1, c\, f^{a+b\, x}\right]}{b\, \text{Log}[f]}$$

- **Program code:**

```
Int[PolyLog[n_,c_.*f_^(a_.+b_.*x_)],x_Symbol] :=
  PolyLog[n+1,c*f^(a+b*x)]/(b*Log[f]) /;
FreeQ[{a,b,c,n},x]
```

$$\int x^m \text{PolyLog}\left[n, c f^{a+b x}\right] dx$$

- **Derivation: Integration by parts**
- **Rule: If $m > 0$, then**

$$\int x^m \text{PolyLog}\left[n, c f^{a+b x}\right] dx \longrightarrow \frac{x^m \text{PolyLog}\left[n+1, c f^{a+b x}\right]}{b \text{Log}[f]} - \frac{m}{b \text{Log}[f]} \int x^{m-1} \text{PolyLog}\left[n+1, c f^{a+b x}\right] dx$$

- **Program code:**

```
Int[x_^m_.*PolyLog[n_,c_.*f_^(a_.+b_.*x_)],x_Symbol] :=
  x^m*PolyLog[n+1,c*f^(a+b*x)]/(b*Log[f]) -
  Dist[m/(b*Log[f]),Int[x^(m-1)*PolyLog[n+1,c*f^(a+b*x)],x]] /;
FreeQ[{a,b,c,n},x] && RationalQ[m] && m>0
```

$$\int \frac{\text{Log}[a\,x^n]^p\,\text{PolyLog}[q,\,b\,x^m]}{x}\,dx$$

- **Derivation: Integration by parts**
- **Rule: If $p > 0$, then**

$$\int \frac{\text{Log}[a\,x^n]^p\,\text{PolyLog}[q,\,b\,x^m]}{x}\,dx \longrightarrow \frac{\text{Log}[a\,x^n]^p\,\text{PolyLog}[q+1,\,b\,x^m]}{m} - \frac{n\,p}{m}\int \frac{\text{Log}[a\,x^n]^{p-1}\,\text{PolyLog}[q+1,\,b\,x^m]}{x}\,dx$$

- **Program code:**

```
Int[Log[a_.*x_^n_.]^p_.*PolyLog[q_,b_.*x_^m_.]/x_,x_Symbol] :=
  Log[a*x^n]^p*PolyLog[q+1,b*x^m]/m -
  Dist[n*p/m,Int[Log[a*x^n]^(p-1)*PolyLog[q+1,b*x^m]/x,x]] /;
FreeQ[{a,b,m,n,q},x] && RationalQ[p] && p>0
```