$\int \text{Tan}[a+b\,x]^n\,dx$

- **Reference: G&R 2.526.17, CRC 292, A&S 4.3.115**
- **Derivation: Reciprocal rule**
- **Basis:** $\text{Tan}[z] = \frac{\text{Sin}[z]}{\text{Cos}[z]}$
- **Rule:**

$$\int \text{Tan}[a+b\,x]\,dx \longrightarrow -\frac{\text{Log}[\text{Cos}[a+b\,x]]}{b}$$

- **Program code:**

```
Int[Tan[a_.+b_.*x_],x_Symbol] :=
  -Log[Cos[a+b*x]]/b /;
FreeQ[{a,b},x]
```

- **Reference: G&R 2.526.33, CRC 293, A&S 4.3.118**

```
Int[Cot[a_.+b_.*x_],x_Symbol] :=
  Log[Sin[a+b*x]]/b /;
FreeQ[{a,b},x]
```

- **Reference: G&R 2.526.22, CRC 420**
- **Derivation: Algebraic expansion**
- **Basis:** $\text{Tan}[z]^2 = -1 + \text{Sec}[z]^2$
- **Rule:**

$$\int \text{Tan}[a+b\,x]^2\,dx \longrightarrow -x + \frac{\text{Tan}[a+b\,x]}{b}$$

- **Program code:**

```
Int[Tan[a_.+b_.*x_]^2,x_Symbol] :=
  -x + Tan[a+b*x]/b /;
FreeQ[{a,b},x]
```

- **Reference: G&R 2.526.38, CRC 424**

```
Int[Cot[a_.+b_.*x_]^2,x_Symbol] :=
  -x - Cot[a+b*x]/b /;
FreeQ[{a,b},x]
```

- **Reference: G&R 2.510.1, CRC 423, A&S 4.3.129**
- **Derivation: Integration by parts with a double-back flip**
- **Basis:** $\text{Tan}[z]^n = \frac{\text{Tan}[z]^{n-1}\,\text{Sin}[z]}{\text{Cos}[z]}$
- **Rule: If** $n > 1$**, then**

$$\int (c\,\text{Tan}[a+b\,x])^n\,dx \longrightarrow \frac{c\,(c\,\text{Tan}[a+b\,x])^{n-1}}{b\,(n-1)} - c^2\int (c\,\text{Tan}[a+b\,x])^{n-2}\,dx$$

- **Program code:**

```
Int[(c_.*Tan[a_.+b_.*x_])^n_,x_Symbol] :=
  c*(c*Tan[a+b*x])^(n-1)/(b*(n-1)) -
  Dist[c^2,Int[(c*Tan[a+b*x])^(n-2),x]] /;
FreeQ[{a,b,c},x] && RationalQ[n] && n>1
```

- **Reference: G&R 2.510.4, CRC 427, A&S 4.3.130**

```
Int[(c_.*Cot[a_.+b_.*x_])^n_,x_Symbol] :=
  -c*(c*Cot[a+b*x])^(n-1)/(b*(n-1)) -
  Dist[c^2,Int[(c*Cot[a+b*x])^(n-2),x]] /;
FreeQ[{a,b,c},x] && RationalQ[n] && n>1
```

- **Reference: G&R 2.510.4, CRC 427'**
- **Derivation: Inverted integration by parts with a double-back flip**
- **Basis:** $\text{Tan}[z]^n = \frac{\text{Tan}[z]^{n-1}\,\text{Sin}[z]}{\text{Cos}[z]}$
- **Rule: If** $n < -1$**, then**

$$\int (c\,\text{Tan}[a+b\,x])^n\,dx \longrightarrow \frac{(c\,\text{Tan}[a+b\,x])^{n+1}}{b\,c\,(n+1)} - \frac{1}{c^2}\int (c\,\text{Tan}[a+b\,x])^{n+2}\,dx$$

- **Program code:**

```
Int[(c_.*Tan[a_.+b_.*x_])^n_,x_Symbol] :=
  (c*Tan[a+b*x])^(n+1)/(b*c*(n+1)) -
  Dist[1/c^2,Int[(c*Tan[a+b*x])^(n+2),x]] /;
FreeQ[{a,b,c},x] && RationalQ[n] && n<-1
```

- **Reference: G&R 2.510.1, CRC 423'**

```
Int[(c_.*Cot[a_.+b_.*x_])^n_,x_Symbol] :=
  -(c*Cot[a+b*x])^(n+1)/(b*c*(n+1)) -
  Dist[1/c^2,Int[(c*Cot[a+b*x])^(n+2),x]] /;
FreeQ[{a,b,c},x] && RationalQ[n] && n<-1
```

$\int (a + b\,\mathrm{Tan}[c + d\,x])^n \,dx$ when $a^2 + b^2 = 0$

- **Rule: If $a^2 + b^2 = 0$, then**

$$\int \frac{1}{a + b\,\mathrm{Tan}[c + d\,x]}\,dx \longrightarrow \frac{x}{2\,a} - \frac{a}{2\,b\,d\,(a + b\,\mathrm{Tan}[c + d\,x])}$$

- **Program code:**

```
Int[1/(a_+b_.*Tan[c_.+d_.*x_]),x_Symbol] :=
  x/(2*a) - a/(2*b*d*(a+b*Tan[c+d*x])) /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2+b^2]
```

```
Int[1/(a_+b_.*Cot[c_.+d_.*x_]),x_Symbol] :=
  x/(2*a) + a/(2*b*d*(a+b*Cot[c+d*x])) /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2+b^2]
```

- **Rule: If $a^2 + b^2 = 0 \land a > 0$, then**

$$\int \sqrt{a + b\,\mathrm{Tan}[c + d\,x]}\,dx \longrightarrow -\frac{\sqrt{2}\,b}{d\,\sqrt{a}}\,\mathrm{ArcTanh}\left[\frac{\sqrt{a + b\,\mathrm{Tan}[c + d\,x]}}{\sqrt{2}\,\sqrt{a}}\right]$$

- **Program code:**

```
Int[Sqrt[a_+b_.*Tan[c_.+d_.*x_]],x_Symbol] :=
  -Sqrt[2]*b*ArcTanh[Sqrt[a+b*Tan[c+d*x]]/(Sqrt[2]*Rt[a,2])]/(d*Rt[a,2]) /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2+b^2] && PosQ[a]
```

```
Int[Sqrt[a_+b_.*Cot[c_.+d_.*x_]],x_Symbol] :=
  Sqrt[2]*b*ArcCoth[Sqrt[a+b*Cot[c+d*x]]/(Sqrt[2]*Rt[a,2])]/(d*Rt[a,2]) /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2+b^2] && PosQ[a]
```

- **Rule: If $a^2 + b^2 = 0 \land \neg (a > 0)$, then**

$$\int \sqrt{a + b\,\mathrm{Tan}[c + d\,x]}\,dx \longrightarrow \frac{\sqrt{2}\,b}{d\,\sqrt{-a}}\,\mathrm{ArcTan}\left[\frac{\sqrt{a + b\,\mathrm{Tan}[c + d\,x]}}{\sqrt{2}\,\sqrt{-a}}\right]$$

- **Program code:**

```
Int[Sqrt[a_+b_.*Tan[c_.+d_.*x_]],x_Symbol] :=
  Sqrt[2]*b*ArcTan[Sqrt[a+b*Tan[c+d*x]]/(Sqrt[2]*Rt[-a,2])]/(d*Rt[-a,2]) /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2+b^2] && NegQ[a]
```

```
Int[Sqrt[a_+b_.*Cot[c_.+d_.*x_]],x_Symbol] :=
  Sqrt[2]*b*ArcCot[Sqrt[a+b*Cot[c+d*x]]/(Sqrt[2]*Rt[-a,2])]/(d*Rt[-a,2]) /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2+b^2] && NegQ[a]
```

- **Rule: If** $a^2 + b^2 = 0 \bigwedge n \in \mathbb{F} \bigwedge n > 1$**, then**

$$\int (a + b \operatorname{Tan}[c + d x])^n \, dx \longrightarrow -\frac{a^2 (a + b \operatorname{Tan}[c + d x])^{n-1}}{b d (n-1)} + 2 a \int (a + b \operatorname{Tan}[c + d x])^{n-1} \, dx$$

- **Program code:**

```
Int[(a_+b_.*Tan[c_.+d_.*x_])^n_,x_Symbol] :=
  -a^2*(a+b*Tan[c+d*x])^(n-1)/(b*d*(n-1)) +
  Dist[2*a,Int[(a+b*Tan[c+d*x])^(n-1),x]] /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2+b^2] && FractionQ[n] && n>1
```

```
Int[(a_+b_.*Cot[c_.+d_.*x_])^n_,x_Symbol] :=
  a^2*(a+b*Cot[c+d*x])^(n-1)/(b*d*(n-1)) +
  Dist[2*a,Int[(a+b*Cot[c+d*x])^(n-1),x]] /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2+b^2] && FractionQ[n] && n>1
```

- **Rule: If** $a^2 + b^2 = 0 \bigwedge n < 0$**, then**

$$\int (a + b \operatorname{Tan}[c + d x])^n \, dx \longrightarrow \frac{a (a + b \operatorname{Tan}[c + d x])^n}{2 b d n} + \frac{1}{2 a} \int (a + b \operatorname{Tan}[c + d x])^{n+1} \, dx$$

- **Program code:**

```
Int[(a_+b_.*Tan[c_.+d_.*x_])^n_,x_Symbol] :=
  a*(a+b*Tan[c+d*x])^n/(2*b*d*n) +
  Dist[1/(2*a),Int[(a+b*Tan[c+d*x])^(n+1),x]] /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2+b^2] && RationalQ[n] && n<0
```

```
Int[(a_+b_.*Cot[c_.+d_.*x_])^n_,x_Symbol] :=
  -a*(a+b*Cot[c+d*x])^n/(2*b*d*n) +
  Dist[1/(2*a),Int[(a+b*Cot[c+d*x])^(n+1),x]] /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2+b^2] && RationalQ[n] && n<0
```

$\int (a + b\,\mathrm{Tan}[c + d\,x])^n\,dx$ when $a^2 + b^2 \neq 0$

- **Derivation: Algebraic expansion and integration by substitution**
- **Basis:** $\frac{1}{a+b\,\mathrm{Tan}[z]} = \frac{a}{a^2+b^2} + \frac{b}{\left(a^2+b^2\right)\,(a\,\mathrm{Cos}[z]+b\,\mathrm{Sin}[z])}\,\partial_z\,(a\,\mathrm{Cos}[z] + b\,\mathrm{Sin}[z])$
- **Rule: If** $a^2 + b^2 \neq 0$**, then**

$$\int \frac{1}{a + b\,\mathrm{Tan}[c + d\,x]}\,dx \longrightarrow \frac{a\,x}{a^2 + b^2} + \frac{b\,\mathrm{Log}[a\,\mathrm{Cos}[c + d\,x] + b\,\mathrm{Sin}[c + d\,x]]}{d\,\left(a^2 + b^2\right)}$$

- **Program code:**

```
Int[1/(a_+b_.*Tan[c_.+d_.*x_]),x_Symbol] :=
  a*x/(a^2+b^2) + b*Log[a*Cos[c+d*x]+b*Sin[c+d*x]]/(d*(a^2+b^2)) /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2+b^2]
```

```
Int[1/(a_+b_.*Cot[c_.+d_.*x_]),x_Symbol] :=
  a*x/(a^2+b^2) - b*Log[b*Cos[c+d*x]+a*Sin[c+d*x]]/(d*(a^2+b^2)) /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2+b^2]
```

- **Derivation: Algebraic expansion**
- **Basis:** $a + b\,z = \frac{a-b\,i}{2}\,(1 + i\,z) + \frac{a+b\,i}{2}\,(1 - i\,z)$
- **Note: Although the resulting integrands are more complicated than the original, they are of the form required for rules in the next section.**
- **Rule: If** $a^2 + b^2 \neq 0$**, then**

$$\int \sqrt{a + b\,\mathrm{Tan}[c + d\,x]}\,dx \longrightarrow \frac{a - b\,i}{2} \int \frac{1 + i\,\mathrm{Tan}[c + d\,x]}{\sqrt{a + b\,\mathrm{Tan}[c + d\,x]}}\,dx + \frac{a + b\,i}{2} \int \frac{1 - i\,\mathrm{Tan}[c + d\,x]}{\sqrt{a + b\,\mathrm{Tan}[c + d\,x]}}\,dx$$

- **Program code:**

```
Int[Sqrt[a_+b_.*Tan[c_.+d_.*x_]],x_Symbol] :=
  Dist[(a-b*I)/2,Int[(1+I*Tan[c+d*x])/Sqrt[a+b*Tan[c+d*x]],x]] +
  Dist[(a+b*I)/2,Int[(1-I*Tan[c+d*x])/Sqrt[a+b*Tan[c+d*x]],x]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2+b^2]
```

```
Int[Sqrt[a_+b_.*Cot[c_.+d_.*x_]],x_Symbol] :=
  Dist[(a-b*I)/2,Int[(1+I*Cot[c+d*x])/Sqrt[a+b*Cot[c+d*x]],x]] +
  Dist[(a+b*I)/2,Int[(1-I*Cot[c+d*x])/Sqrt[a+b*Cot[c+d*x]],x]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2+b^2]
```

- **Derivation: Algebraic expansion**
- **Basis:** $1 = \frac{1}{2}(1 + i z) + \frac{1}{2}(1 - i z)$
- **Note: Although the resulting integrands are more complicated than the original, they are of the form required for rules in the next section.**
- **Rule: If** $a^2 + b^2 \neq 0$**, then**

$$\int \frac{1}{\sqrt{a + b \operatorname{Tan}[c + d x]}} \, dx \longrightarrow \frac{1}{2} \int \frac{1 + i \operatorname{Tan}[c + d x]}{\sqrt{a + b \operatorname{Tan}[c + d x]}} \, dx + \frac{1}{2} \int \frac{1 - i \operatorname{Tan}[c + d x]}{\sqrt{a + b \operatorname{Tan}[c + d x]}} \, dx$$

- **Program code:**

```
Int[1/Sqrt[a_+b_.*Tan[c_.+d_.*x_]],x_Symbol] :=
  Dist[1/2,Int[(1+I*Tan[c+d*x])/Sqrt[a+b*Tan[c+d*x]],x]] +
  Dist[1/2,Int[(1-I*Tan[c+d*x])/Sqrt[a+b*Tan[c+d*x]],x]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2+b^2]
```

```
Int[1/Sqrt[a_+b_.*Cot[c_.+d_.*x_]],x_Symbol] :=
  Dist[1/2,Int[(1+I*Cot[c+d*x])/Sqrt[a+b*Cot[c+d*x]],x]] +
  Dist[1/2,Int[(1-I*Cot[c+d*x])/Sqrt[a+b*Cot[c+d*x]],x]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2+b^2]
```

- **Rule: If** $a^2 + b^2 \neq 0 \land n > 1$**, then**

$$\int (a + b \operatorname{Tan}[c + d x])^n \, dx \longrightarrow$$

$$\frac{b (a + b \operatorname{Tan}[c + d x])^{n-1}}{d (n - 1)} + \int \left(a^2 - b^2 + 2 a b \operatorname{Tan}[c + d x]\right) (a + b \operatorname{Tan}[c + d x])^{n-2} \, dx$$

- **Program code:**

```
Int[(a_+b_.*Tan[c_.+d_.*x_])^n_,x_Symbol] :=
  b*(a+b*Tan[c+d*x])^(n-1)/(d*(n-1)) +
  Int[(a^2-b^2+2*a*b*Tan[c+d*x])*(a+b*Tan[c+d*x])^(n-2),x] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2+b^2] && FractionQ[n] && n>1
```

```
Int[(a_+b_.*Cot[c_.+d_.*x_])^n_,x_Symbol] :=
  -b*(a+b*Cot[c+d*x])^(n-1)/(d*(n-1)) +
  Int[(a^2-b^2+2*a*b*Cot[c+d*x])*(a+b*Cot[c+d*x])^(n-2),x] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2+b^2] && FractionQ[n] && n>1
```

- **Rule: If** $a^2+b^2\neq 0\land n<-1$**, then**

$$\int (a+b\,\mathrm{Tan}[c+d\,x])^n\,dx \longrightarrow$$

$$\frac{b\,(a+b\,\mathrm{Tan}[c+d\,x])^{n+1}}{d\,\left(a^2+b^2\right)\,(n+1)}+\frac{1}{a^2+b^2}\int (a-b\,\mathrm{Tan}[c+d\,x])\,(a+b\,\mathrm{Tan}[c+d\,x])^{n+1}\,dx$$

- **Program code:**

```
Int[(a_+b_.*Tan[c_.+d_.*x_])^n_,x_Symbol] :=
  (b*(a+b*Tan[c+d*x])^(n+1))/(d*(a^2+b^2)*(n+1)) +
  Dist[1/(a^2+b^2),Int[(a-b*Tan[c+d*x])*(a+b*Tan[c+d*x])^(n+1),x]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2+b^2] && RationalQ[n] && n<-1
```

```
Int[(a_+b_.*Cot[c_.+d_.*x_])^n_,x_Symbol] :=
  -(b*(a+b*Cot[c+d*x])^(n+1))/(d*(a^2+b^2)*(n+1)) +
  Dist[1/(a^2+b^2),Int[(a-b*Cot[c+d*x])*(a+b*Cot[c+d*x])^(n+1),x]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2+b^2] && RationalQ[n] && n<-1
```

$$\int (A + B \operatorname{Tan}[c + d x]) (a + b \operatorname{Tan}[c + d x])^n \, dx$$

- **Note: None of these rules appear in published integral tables.**

- **Derivation: Algebraic expansion**

- **Basis:** $\frac{A+B z}{a+b z} = \frac{B}{b} + \frac{b A-a B}{b} \frac{1}{a+b z}$

- **Rule: If** $b A - a B \neq 0$**, then**

$$\int \frac{A + B \operatorname{Tan}[c + d x]}{a + b \operatorname{Tan}[c + d x]} dx \longrightarrow \frac{B x}{b} + \frac{b A - a B}{b} \int \frac{1}{a + b \operatorname{Tan}[c + d x]} dx$$

- **Program code:**

```
Int[(A_.+B_.*Tan[c_.+d_.*x_])/(a_.+b_.*Tan[c_.+d_.*x_]),x_Symbol] :=
  B*x/b + Dist[(b*A-a*B)/b,Int[1/(a+b*Tan[c+d*x]),x]] /;
FreeQ[{a,b,c,d,A,B},x] && NonzeroQ[b*A-a*B]
```

```
Int[(A_.+B_.*Cot[c_.+d_.*x_])/(a_.+b_.*Cot[c_.+d_.*x_]),x_Symbol] :=
  B*x/b + Dist[(b*A-a*B)/b,Int[1/(a+b*Cot[c+d*x]),x]] /;
FreeQ[{a,b,c,d,A,B},x] && NonzeroQ[b*A-a*B]
```

- **Rule: If** $A^2 + B^2 = 0 \land b A + a B \neq 0$**, then**

$$\int \frac{A + B \operatorname{Tan}[c + d x]}{\sqrt{a + b \operatorname{Tan}[c + d x]}} dx \longrightarrow -\frac{2 B}{d \sqrt{a + \frac{b A}{B}}} \operatorname{ArcTanh}\left[\frac{\sqrt{a + b \operatorname{Tan}[c + d x]}}{\sqrt{a + \frac{b A}{B}}}\right]$$

- **Program code:**

```
Int[(A_+B_.*Tan[c_.+d_.*x_])/Sqrt[a_.+b_.*Tan[c_.+d_.*x_]],x_Symbol] :=
  -2*B*ArcTanh[Sqrt[a+b*Tan[c+d*x]]/Rt[a+b*A/B,2]]/(d*Rt[a+b*A/B,2]) /;
FreeQ[{a,b,c,d,A,B},x] && ZeroQ[A^2+B^2] && NonzeroQ[b*A+a*B]
```

```
Int[(A_+B_.*Cot[c_.+d_.*x_])/Sqrt[a_.+b_.*Cot[c_.+d_.*x_]],x_Symbol] :=
  2*B*ArcCoth[Sqrt[a+b*Cot[c+d*x]]/Rt[a+b*A/B,2]]/(d*Rt[a+b*A/B,2]) /;
FreeQ[{a,b,c,d,A,B},x] && ZeroQ[A^2+B^2] && NonzeroQ[b*A+a*B]
```

- **Derivation: Algebraic expansion**

- **Basis:** $A + B z = \frac{A - B i}{2} (1 + i z) + \frac{A + B i}{2} (1 - i z)$

- **Rule: If** $A^2 + B^2 \neq 0 \wedge a^2 + b^2 \neq 0$**, then**

$$\int \frac{A + B \operatorname{Tan}[c + d x]}{\sqrt{a + b \operatorname{Tan}[c + d x]}} \, dx \longrightarrow \frac{A - B i}{2} \int \frac{1 + i \operatorname{Tan}[c + d x]}{\sqrt{a + b \operatorname{Tan}[c + d x]}} \, dx + \frac{A + B i}{2} \int \frac{1 - i \operatorname{Tan}[c + d x]}{\sqrt{a + b \operatorname{Tan}[c + d x]}} \, dx$$

- **Program code:**

```
Int[(A_.+B_.*Tan[c_.+d_.*x_])/Sqrt[a_.+b_.*Tan[c_.+d_.*x_]],x_Symbol] :=
  Dist[(A-B*I)/2,Int[(1+I*Tan[c+d*x])/Sqrt[a+b*Tan[c+d*x]],x]] +
  Dist[(A+B*I)/2,Int[(1-I*Tan[c+d*x])/Sqrt[a+b*Tan[c+d*x]],x]] /;
FreeQ[{a,b,c,d,A,B},x] && NonzeroQ[A^2+B^2] && NonzeroQ[a^2+b^2]
```

```
Int[(A_.+B_.*Cot[c_.+d_.*x_])/Sqrt[a_.+b_.*Cot[c_.+d_.*x_]],x_Symbol] :=
  Dist[(A-B*I)/2,Int[(1+I*Cot[c+d*x])/Sqrt[a+b*Cot[c+d*x]],x]] +
  Dist[(A+B*I)/2,Int[(1-I*Cot[c+d*x])/Sqrt[a+b*Cot[c+d*x]],x]] /;
FreeQ[{a,b,c,d,A,B},x] && NonzeroQ[A^2+B^2] && NonzeroQ[a^2+b^2]
```

- **Rule: If** $n \in \mathbb{F} \wedge n > 0 \wedge b A + a B = 0$**, then**

$$\int (A + B \operatorname{Tan}[c + d x]) \, (a + b \operatorname{Tan}[c + d x])^n \, dx \longrightarrow$$
$$\frac{B \, (a + b \operatorname{Tan}[c + d x])^n}{d n} + (a A - b B) \int (a + b \operatorname{Tan}[c + d x])^{n-1} \, dx$$

- **Program code:**

```
Int[(A_.+B_.*Tan[c_.+d_.*x_])*(a_+b_.*Tan[c_.+d_.*x_])^n_,x_Symbol] :=
  B*(a+b*Tan[c+d*x])^n/(d*n) +
  Dist[a*A-b*B,Int[(a+b*Tan[c+d*x])^(n-1),x]] /;
FreeQ[{a,b,c,d,A,B},x] && FractionQ[n] && n>0 && ZeroQ[b*A+a*B]
```

```
Int[(A_.+B_.*Cot[c_.+d_.*x_])*(a_+b_.*Cot[c_.+d_.*x_])^n_,x_Symbol] :=
  -B*(a+b*Cot[c+d*x])^n/(d*n) +
  Dist[a*A-b*B,Int[(a+b*Cot[c+d*x])^(n-1),x]] /;
FreeQ[{a,b,c,d,A,B},x] && FractionQ[n] && n>0 && ZeroQ[b*A+a*B]
```

- **Rule: If** $n \in \mathbb{F} \wedge n > 0 \wedge b A + a B \neq 0$, **then**

$$\int (A + B \operatorname{Tan}[c + d x]) (a + b \operatorname{Tan}[c + d x])^n \, dx \longrightarrow \frac{B (a + b \operatorname{Tan}[c + d x])^n}{d n} + \int (a A - b B + (b A + a B) \operatorname{Tan}[c + d x]) (a + b \operatorname{Tan}[c + d x])^{n-1} \, dx$$

- **Program code:**

```
Int[(A_.+B_.*Tan[c_.+d_.*x_])*(a_+b_.*Tan[c_.+d_.*x_])^n_,x_Symbol] :=
  B*(a+b*Tan[c+d*x])^n/(d*n) +
  Int[(a*A-b*B+(b*A+a*B)*Tan[c+d*x])*(a+b*Tan[c+d*x])^(n-1),x] /;
FreeQ[{a,b,c,d,A,B},x] && FractionQ[n] && n>0 && NonzeroQ[b*A+a*B]
```

```
Int[(A_.+B_.*Cot[c_.+d_.*x_])*(a_+b_.*Cot[c_.+d_.*x_])^n_,x_Symbol] :=
  -B*(a+b*Cot[c+d*x])^n/(d*n) +
  Int[(a*A-b*B+(b*A+a*B)*Cot[c+d*x])*(a+b*Cot[c+d*x])^(n-1),x] /;
FreeQ[{a,b,c,d,A,B},x] && FractionQ[n] && n>0 && NonzeroQ[b*A+a*B]
```

- **Rule: If** $a^2 + b^2 \neq 0 \wedge n < -1 \wedge b A - a B \neq 0$, **then**

$$\int (A + B \operatorname{Tan}[c + d x]) (a + b \operatorname{Tan}[c + d x])^n \, dx \longrightarrow \frac{(b A - a B) (a + b \operatorname{Tan}[c + d x])^{n+1}}{d (a^2 + b^2) (n + 1)} + \frac{1}{a^2 + b^2} \int (a A + b B - (b A - a B) \operatorname{Tan}[c + d x]) (a + b \operatorname{Tan}[c + d x])^{n+1} \, dx$$

- **Program code:**

```
Int[(A_.+B_.*Tan[c_.+d_.*x_])*(a_+b_.*Tan[c_.+d_.*x_])^n_,x_Symbol] :=
  (b*A-a*B)*(a+b*Tan[c+d*x])^(n+1)/(d*(a^2+b^2)*(n+1)) +
  Dist[1/(a^2+b^2),Int[(a*A+b*B-(b*A-a*B)*Tan[c+d*x])*(a+b*Tan[c+d*x])^(n+1),x]] /;
FreeQ[{a,b,c,d,A,B},x] && NonzeroQ[a^2+b^2] && RationalQ[n] && n<-1 && NonzeroQ[b*A-a*B]
```

```
Int[(A_.+B_.*Cot[c_.+d_.*x_])*(a_+b_.*Cot[c_.+d_.*x_])^n_,x_Symbol] :=
  -(b*A-a*B)*(a+b*Cot[c+d*x])^(n+1)/(d*(a^2+b^2)*(n+1)) +
  Dist[1/(a^2+b^2),Int[(a*A+b*B-(b*A-a*B)*Cot[c+d*x])*(a+b*Cot[c+d*x])^(n+1),x]] /;
FreeQ[{a,b,c,d,A,B},x] && NonzeroQ[a^2+b^2] && RationalQ[n] && n<-1 && NonzeroQ[b*A-a*B]
```

$$\int \left(a+b\,\mathrm{Tan}[c+d\,x]^2\right)^n dx$$

- **Derivation: Algebraic simplification**
- **Basis: If** $a-b=0$**, then** $a+b\,\mathrm{Tan}[z]^2=b\,\mathrm{Sec}[z]^2$
- **Rule: If** $a-b=0\land m\in\mathbb{Z}$**, then**

$$\int u\left(a+b\,\mathrm{Tan}[v]^2\right)^m dx\;\longrightarrow\;b^m\int u\,\mathrm{Sec}[v]^{2m}\,dx$$

- **Program code:**

```
Int[u_.*(a_+b_.*Tan[v_]^2)^m_,x_Symbol] :=
  Dist[b^m,Int[u*Sec[v]^(2*m),x]] /;
FreeQ[{a,b,m},x] && ZeroQ[a-b] && IntegerQ[m]
```

```
Int[u_.*(a_+b_.*Cot[v_]^2)^m_,x_Symbol] :=
  Dist[b^m,Int[u*Csc[v]^(2*m),x]] /;
FreeQ[{a,b,m},x] && ZeroQ[a-b] && IntegerQ[m]
```

- **Derivation: Algebraic simplification**
- **Basis: If** $a-b=0$**, then** $a+b\,\mathrm{Tan}[z]^2=b\,\mathrm{Sec}[z]^2$
- **Rule: If** $a-b=0\land m\notin\mathbb{Z}$**, then**

$$\int u\left(a+b\,\mathrm{Tan}[v]^2\right)^m dx\;\longrightarrow\;\int u\left(b\,\mathrm{Sec}[v]^2\right)^m dx$$

- **Program code:**

```
Int[u_.*(a_+b_.*Tan[v_]^2)^m_,x_Symbol] :=
  Int[u*(b*Sec[v]^2)^m,x] /;
FreeQ[{a,b,m},x] && ZeroQ[a-b] && Not[IntegerQ[m]]
```

```
Int[u_.*(a_+b_.*Cot[v_]^2)^m_,x_Symbol] :=
  Int[u*(b*Csc[v]^2)^m,x] /;
FreeQ[{a,b,m},x] && ZeroQ[a-b] && Not[IntegerQ[m]]
```

- **Rule: If $a - b \neq 0$, then**

$$\int \frac{1}{a + b \operatorname{Tan}[c + d x]^2} \, dx \longrightarrow \frac{x}{a - b} - \frac{\sqrt{b}}{\sqrt{a} \, d \, (a - b)} \operatorname{ArcTan}\left[\frac{\sqrt{b} \operatorname{Tan}[c + d x]}{\sqrt{a}}\right]$$

- **Program code:**

```
Int[1/(a_+b_.*Tan[c_.+d_.*x_]^2),x_Symbol] :=
  x/(a-b) - Sqrt[b]*ArcTan[Sqrt[b]*Tan[c+d*x]/Sqrt[a]]/(Sqrt[a]*d*(a-b)) /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a-b]
```

```
Int[1/(a_+b_.*Cot[c_.+d_.*x_]^2),x_Symbol] :=
  x/(a-b) + Sqrt[b]*ArcTan[Sqrt[b]*Cot[c+d*x]/Sqrt[a]]/(Sqrt[a]*d*(a-b)) /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a-b]
```

$$\int x^m \,\mathrm{Tan}[a + b\, x^n]^p \,dx$$

- **Derivation: Algebraic expansion**
- **Basis:** $\mathrm{Tan}[z] = -i + \frac{2 i}{1+e^{2 i z}}$
- **Rule: If** $m \in \mathbb{Z} \wedge m > 0$**, then**

$$\int x^m \,\mathrm{Tan}[a + b\, x] \,dx \longrightarrow -\frac{i\, x^{m+1}}{m+1} + 2 i \int \frac{x^m}{1 + e^{2 i a + 2 i b x}} \,dx$$

- **Program code:**

```
Int[x_^m_.*Tan[a_.+b_.*x_^n_.],x_Symbol] :=
  -I*x^(m+1)/(m+1) +
  Dist[2*I,Int[x^m/(1+E^(2*I*a+2*I*b*x^n)),x]] /;
FreeQ[{a,b,n},x] && IntegerQ[m] && m>0 && NonzeroQ[m-n+1] && n===1
```

```
Int[x_^m_.*Cot[a_.+b_.*x_^n_.],x_Symbol] :=
  I*x^(m+1)/(m+1) -
  Dist[2*I,Int[x^m/(1-E^(2*I*a+2*I*b*x^n)),x]] /;
FreeQ[{a,b,n},x] && IntegerQ[m] && m>0 && NonzeroQ[m-n+1] && n===1
```

- **Note: This rule does not appear in published integral tables.**
- **Rule: If** $p > 1 \wedge m - n + 1 \neq 0 \wedge 0 < n \leq m$**, then**

$$\int x^m \,\mathrm{Tan}[a + b\, x^n]^p \,dx \longrightarrow \frac{x^{m-n+1} \,\mathrm{Tan}[a + b\, x^n]^{p-1}}{b\, n\, (p-1)} - \frac{m-n+1}{b\, n\, (p-1)} \int x^{m-n} \,\mathrm{Tan}[a + b\, x^n]^{p-1} \,dx - \int x^m \,\mathrm{Tan}[a + b\, x^n]^{p-2} \,dx$$

- **Program code:**

```
Int[x_^m_.*Tan[a_.+b_.*x_^n_.]^p_,x_Symbol] :=
  x^(m-n+1)*Tan[a+b*x^n]^(p-1)/(b*n*(p-1)) -
  Dist[(m-n+1)/(b*n*(p-1)),Int[x^(m-n)*Tan[a+b*x^n]^(p-1),x]] -
  Int[x^m*Tan[a+b*x^n]^(p-2),x] /;
FreeQ[{a,b},x] && RationalQ[{m,n,p}] && p>1 && NonzeroQ[m-n+1] && 0<n<=m
```

```
Int[x_^m_.*Cot[a_.+b_.*x_^n_.]^p_,x_Symbol] :=
  -x^(m-n+1)*Cot[a+b*x^n]^(p-1)/(b*n*(p-1)) +
  Dist[(m-n+1)/(b*n*(p-1)),Int[x^(m-n)*Cot[a+b*x^n]^(p-1),x]] -
  Int[x^m*Cot[a+b*x^n]^(p-2),x] /;
FreeQ[{a,b},x] && RationalQ[{m,n,p}] && p>1 && NonzeroQ[m-n+1] && 0<n<=m
```

$$\int x^m \operatorname{Tan}\left[a + b x + c x^2\right] dx$$

- **Rule:**

$$\int x \operatorname{Tan}\left[a + b x + c x^2\right] dx \longrightarrow -\frac{\operatorname{Log}\left[\operatorname{Cos}\left[a + b x + c x^2\right]\right]}{2 c} - \frac{b}{2 c} \int \operatorname{Tan}\left[a + b x + c x^2\right] dx$$

- **Program code:**

```
Int[x_*Tan[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  -Log[Cos[a+b*x+c*x^2]]/(2*c) -
  Dist[b/(2*c),Int[Tan[a+b*x+c*x^2],x]] /;
FreeQ[{a,b,c},x]
```

```
Int[x_*Cot[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  Log[Sin[a+b*x+c*x^2]]/(2*c) -
  Dist[b/(2*c),Int[Cot[a+b*x+c*x^2],x]] /;
FreeQ[{a,b,c},x]
```

- **Note: This rule is valid, but to be useful need a rule for reducing integrands of the form** $x^m \operatorname{Log}[\operatorname{Cos}[a + b x + c x^2]]$.

- **Rule: If** $m > 1$**, then**

$$\int x^m \operatorname{Tan}\left[a + b x + c x^2\right] dx \longrightarrow -\frac{x^{m-1} \operatorname{Log}\left[\operatorname{Cos}\left[a + b x + c x^2\right]\right]}{2 c} - \frac{b}{2 c} \int x^{m-1} \operatorname{Tan}\left[a + b x + c x^2\right] dx + \frac{m-1}{2 c} \int x^{m-2} \operatorname{Log}\left[\operatorname{Cos}\left[a + b x + c x^2\right]\right] dx$$

- **Program code:**

```
(* Int[x_^m_*Tan[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  -x^(m-1)*Log[Cos[a+b*x+c*x^2]]/(2*c) -
  Dist[b/(2*c),Int[x^(m-1)*Tan[a+b*x+c*x^2],x]] +
  Dist[(m-1)/(2*c),Int[x^(m-2)*Log[Cos[a+b*x+c*x^2]],x]] /;
FreeQ[{a,b,c},x] && RationalQ[m] && m>1 *)
```

```
(* Int[x_^m_*Cot[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  x^(m-1)*Log[Sin[a+b*x+c*x^2]]/(2*c) -
  Dist[b/(2*c),Int[x^(m-1)*Cot[a+b*x+c*x^2],x]] -
  Dist[(m-1)/(2*c),Int[x^(m-2)*Log[Sin[a+b*x+c*x^2]],x]] /;
FreeQ[{a,b,c},x] && RationalQ[m] && m>1*)
```