$$\int \text{FresnelS}[a+b\,x]^n \, dx$$

- **Derivation: Integration by parts**
- **Rule:**

$$\int \text{FresnelS}[a+b\,x] \, dx \longrightarrow \frac{(a+b\,x)\,\text{FresnelS}[a+b\,x]}{b} + \frac{\text{Cos}\left[\frac{\pi}{2}\,(a+b\,x)^2\right]}{b\,\pi}$$

- **Program code:**

```
Int[FresnelS[a_.+b_.*x_],x_Symbol] :=
  (a+b*x)*FresnelS[a+b*x]/b + Cos[Pi/2*(a+b*x)^2]/(b*Pi) /;
FreeQ[{a,b},x]
```

```
Int[FresnelC[a_.+b_.*x_],x_Symbol] :=
  (a+b*x)*FresnelC[a+b*x]/b - Sin[Pi/2*(a+b*x)^2]/(b*Pi) /;
FreeQ[{a,b},x]
```

- **Derivation: Integration by parts**
- **Rule:**

$$\int \text{FresnelS}[a+b\,x]^2 \, dx \longrightarrow \frac{(a+b\,x)\,\text{FresnelS}[a+b\,x]^2}{b} - 2\int (a+b\,x)\,\text{Sin}\left[\frac{\pi}{2}\,(a+b\,x)^2\right]\text{FresnelS}[a+b\,x] \, dx$$

- **Program code:**

```
Int[FresnelS[a_.+b_.*x_]^2,x_Symbol] :=
  (a+b*x)*FresnelS[a+b*x]^2/b -
  Dist[2,Int[(a+b*x)*Sin[Pi/2*(a+b*x)^2]*FresnelS[a+b*x],x]] /;
FreeQ[{a,b},x]
```

```
Int[FresnelC[a_.+b_.*x_]^2,x_Symbol] :=
  (a+b*x)*FresnelC[a+b*x]^2/b -
  Dist[2,Int[(a+b*x)*Cos[Pi/2*(a+b*x)^2]*FresnelC[a+b*x],x]] /;
FreeQ[{a,b},x]
```

$\int x^m \text{FresnelS}[a + b x]^n \, dx$

- **Derivation: Integration by parts**
- **Rule: If $m + 1 \neq 0$, then**

$$\int x^m \text{FresnelS}[a + b x] \, dx \longrightarrow \frac{x^{m+1} \text{FresnelS}[a + b x]}{m + 1} - \frac{b}{m + 1} \int x^{m+1} \text{Sin}\left[\frac{\pi}{2} (a + b x)^2\right] dx$$

- **Program code:**

```
Int[x_^m_.*FresnelS[a_.+b_.*x_],x_Symbol] :=
  x^(m+1)*FresnelS[a+b*x]/(m+1) -
  Dist[b/(m+1),Int[x^(m+1)*Sin[Pi/2*(a+b*x)^2],x]] /;
FreeQ[{a,b,m},x] && NonzeroQ[m+1]
```

```
Int[x_^m_.*FresnelC[a_.+b_.*x_],x_Symbol] :=
  x^(m+1)*FresnelC[a+b*x]/(m+1) -
  Dist[b/(m+1),Int[x^(m+1)*Cos[Pi/2*(a+b*x)^2],x]] /;
FreeQ[{a,b,m},x] && NonzeroQ[m+1]
```

- **Derivation: Integration by parts**
- **Note: Also apply rule when m mod 4 = 1 when a closed-form antiderivative is defined for $\text{Cos}\left[\frac{\pi x^2}{2}\right] \text{FresnelS}[x]$.**
- **Rule: If $m \in \mathbb{Z} \bigwedge m + 1 \neq 0 \bigwedge \left(m > 0 \bigwedge \frac{m}{2} \in \mathbb{Z} \bigvee m \bmod 4 = 3\right)$, then**

$$\int x^m \text{FresnelS}[b x]^2 \, dx \longrightarrow \frac{x^{m+1} \text{FresnelS}[b x]^2}{m + 1} - \frac{2 b}{m + 1} \int x^{m+1} \text{Sin}\left[\frac{\pi}{2} b^2 x^2\right] \text{FresnelS}[b x] \, dx$$

- **Program code:**

```
Int[x_^m_*FresnelS[b_.*x_]^2,x_Symbol] :=
  x^(m+1)*FresnelS[b*x]^2/(m+1) -
  Dist[2*b/(m+1),Int[x^(m+1)*Sin[Pi/2*b^2*x^2]*FresnelS[b*x],x]] /;
FreeQ[b,x] && IntegerQ[m] && m+1!=0 && (m>0 && EvenQ[m] || Mod[m,4]==3)
```

```
Int[x_^m_*FresnelC[b_.*x_]^2,x_Symbol] :=
  x^(m+1)*FresnelC[b*x]^2/(m+1) -
  Dist[2*b/(m+1),Int[x^(m+1)*Cos[Pi/2*b^2*x^2]*FresnelC[b*x],x]] /;
FreeQ[b,x] && IntegerQ[m] && m+1!=0 && (m>0 && EvenQ[m] || Mod[m,4]==3)
```

- **Derivation: Integration by substitution**

- **Basis:** $x^m f[a+b\,x] = \frac{1}{b}\left(-\frac{a}{b}+\frac{a+b\,x}{b}\right)^m f[a+b\,x]\,\partial_x(a+b\,x)$

- **Note: Rule not necessary until a closed-form antiderivative is defined for** $\text{Cos}\left[\frac{\pi\,x^2}{2}\right]\text{FresnelS}[x]$.

- **Rule: If** $m\in\mathbb{Z}\,\wedge\,m>0$, **then**

$$\int x^m\,\text{FresnelS}[a+b\,x]^2\,dx\;\longrightarrow\;\frac{1}{b}\,\text{Subst}\left[\int\left(-\frac{a}{b}+\frac{x}{b}\right)^m\text{FresnelS}[x]^2\,dx,\;x,\;a+b\,x\right]$$

- **Program code:**

```
(* Int[x_^m_.*FresnelS[a_+b_.*x_]^2,x_Symbol] :=
  Dist[1/b,Subst[Int[(-a/b+x/b)^m*FresnelS[x]^2,x],x,a+b*x]] /;
FreeQ[{a,b},x] && IntegerQ[m] && m>0 *)
```

```
(* Int[x_^m_.*FresnelC[a_+b_.*x_]^2,x_Symbol] :=
  Dist[1/b,Subst[Int[(-a/b+x/b)^m*FresnelC[x]^2,x],x,a+b*x]] /;
FreeQ[{a,b},x] && IntegerQ[m] && m>0 *)
```

$$\int x^m \operatorname{Sin}\left[\frac{\pi}{2}\ b^2 x^2\right] \operatorname{FresnelS}[b\,x]\, dx$$

- **Derivation: Integration by parts special case**
- **Rule:**

$$\int x \operatorname{Sin}\left[\frac{\pi}{2}\ b^2 x^2\right] \operatorname{FresnelS}[b\,x]\, dx \longrightarrow -\frac{\operatorname{Cos}\left[\frac{\pi}{2}\ b^2 x^2\right] \operatorname{FresnelS}[b\,x]}{\pi\, b^2} + \frac{1}{2\,\pi\, b} \int \operatorname{Sin}\left[\pi\ b^2 x^2\right] dx$$

- **Program code:**

```
Int[x_*Sin[c_.*x_^2]*FresnelS[b_.*x_],x_Symbol] :=
  -Cos[Pi/2*b^2*x^2]*FresnelS[b*x]/(Pi*b^2) +
  Dist[1/(2*Pi*b),Int[Sin[Pi*b^2*x^2],x]] /;
FreeQ[{b,c},x] && ZeroQ[c-Pi/2*b^2]
```

```
Int[x_*Cos[c_.*x_^2]*FresnelC[b_.*x_],x_Symbol] :=
  Sin[Pi/2*b^2*x^2]*FresnelC[b*x]/(Pi*b^2) -
  Dist[1/(2*Pi*b),Int[Sin[Pi*b^2*x^2],x]] /;
FreeQ[{b,c},x] && ZeroQ[c-Pi/2*b^2]
```

- **Derivation: Integration by parts**
- **Note: Also apply rule when m mod 4 = 2 when a closed-form antiderivative is defined for** $\operatorname{Cos}\left[\frac{\pi x^2}{2}\right] \operatorname{FresnelS}[x]$.
- **Rule: If** $m \in \mathbb{Z} \land m > 1 \land \neg (m \bmod 4 = 2)$, **then**

$$\int x^m \operatorname{Sin}\left[\frac{\pi}{2}\ b^2 x^2\right] \operatorname{FresnelS}[b\,x]\, dx \longrightarrow -\frac{x^{m-1} \operatorname{Cos}\left[\frac{\pi}{2}\ b^2 x^2\right] \operatorname{FresnelS}[b\,x]}{\pi\, b^2} + \frac{1}{2\,\pi\, b} \int x^{m-1} \operatorname{Sin}\left[\pi\ b^2 x^2\right] dx + \frac{m-1}{\pi\, b^2} \int x^{m-2} \operatorname{Cos}\left[\frac{\pi}{2}\ b^2 x^2\right] \operatorname{FresnelS}[b\,x]\, dx$$

- **Program code:**

```
Int[x_^m_*Sin[c_.*x_^2]*FresnelS[b_.*x_],x_Symbol] :=
  -x^(m-1)*Cos[Pi/2*b^2*x^2]*FresnelS[b*x]/(Pi*b^2) +
  Dist[1/(2*Pi*b),Int[x^(m-1)*Sin[Pi*b^2*x^2],x]] +
  Dist[(m-1)/(Pi*b^2),Int[x^(m-2)*Cos[Pi/2*b^2*x^2]*FresnelS[b*x],x]] /;
FreeQ[{b,c},x] && ZeroQ[c-Pi/2*b^2] && IntegerQ[m] && m>1 && Not[Mod[m,4]==2]
```

```
Int[x_^m_*Cos[c_.*x_^2]*FresnelC[b_.*x_],x_Symbol] :=
  x^(m-1)*Sin[Pi/2*b^2*x^2]*FresnelC[b*x]/(Pi*b^2) -
  Dist[1/(2*Pi*b),Int[x^(m-1)*Sin[Pi*b^2*x^2],x]] -
  Dist[(m-1)/(Pi*b^2),Int[x^(m-2)*Sin[Pi/2*b^2*x^2]*FresnelC[b*x],x]] /;
FreeQ[{b,c},x] && ZeroQ[c-Pi/2*b^2] && IntegerQ[m] && m>1 && Not[Mod[m,4]==2]
```

- **Derivation: Inverted integration by parts**

- **Rule: If** $m \in \mathbb{Z} \land m < -2 \land m \bmod 4 = 0$**, then**

$$\int x^m \operatorname{Sin}\left[\frac{\pi}{2}\, b^2 x^2\right] \operatorname{FresnelS}[b\,x]\, dx \longrightarrow \frac{x^{m+1} \operatorname{Sin}\left[\frac{\pi}{2}\, b^2 x^2\right] \operatorname{FresnelS}[b\,x]}{m+1} - \frac{b\,x^{m+2}}{2\,(m+1)\,(m+2)} + \frac{b}{2\,(m+1)} \int x^{m+1} \operatorname{Cos}\left[\pi\, b^2 x^2\right] dx - \frac{\pi\, b^2}{m+1} \int x^{m+2} \operatorname{Cos}\left[\frac{\pi}{2}\, b^2 x^2\right] \operatorname{FresnelS}[b\,x]\, dx$$

- **Program code:**

```
Int[x_^m_*Sin[c_.*x_^2]*FresnelS[b_.*x_],x_Symbol] :=
  x^(m+1)*Sin[Pi/2*b^2*x^2]*FresnelS[b*x]/(m+1) - b*x^(m+2)/(2*(m+1)*(m+2)) +
  Dist[b/(2*(m+1)),Int[x^(m+1)*Cos[Pi*b^2*x^2],x]] -
  Dist[Pi*b^2/(m+1),Int[x^(m+2)*Cos[Pi/2*b^2*x^2]*FresnelS[b*x],x]] /;
FreeQ[{b,c},x] && ZeroQ[c-Pi/2*b^2] && IntegerQ[m] && m<-2 && Mod[m,4]==0
```

```
Int[x_^m_*Cos[c_.*x_^2]*FresnelC[b_.*x_],x_Symbol] :=
  x^(m+1)*Cos[Pi/2*b^2*x^2]*FresnelC[b*x]/(m+1) - b*x^(m+2)/(2*(m+1)*(m+2)) -
  Dist[b/(2*(m+1)),Int[x^(m+1)*Cos[Pi*b^2*x^2],x]] +
  Dist[Pi*b^2/(m+1),Int[x^(m+2)*Sin[Pi/2*b^2*x^2]*FresnelC[b*x],x]] /;
FreeQ[{b,c},x] && ZeroQ[c-Pi/2*b^2] && IntegerQ[m] && m<-2 && Mod[m,4]==0
```

$$\int x^m \operatorname{Cos}\left[\frac{\pi}{2}\ b^2 x^2\right] \operatorname{FresnelS}[b\,x]\, dx$$

- **Derivation: Integration by parts special case**
- **Rule:**

$$\int x \operatorname{Cos}\left[\frac{\pi}{2}\ b^2 x^2\right] \operatorname{FresnelS}[b\,x]\, dx \longrightarrow \frac{\operatorname{Sin}\left[\frac{\pi}{2}\ b^2 x^2\right] \operatorname{FresnelS}[b\,x]}{\pi\, b^2} - \frac{x}{2\,\pi\, b} + \frac{1}{2\,\pi\, b} \int \operatorname{Cos}\left[\pi\ b^2 x^2\right] dx$$

- **Program code:**

```
Int[x_*Cos[c_.*x_^2]*FresnelS[b_.*x_],x_Symbol] :=
  Sin[Pi/2*b^2*x^2]*FresnelS[b*x]/(Pi*b^2) - x/(2*Pi*b) +
  Dist[1/(2*Pi*b),Int[Cos[Pi*b^2*x^2],x]] /;
FreeQ[{b,c},x] && ZeroQ[c-Pi/2*b^2]
```

```
Int[x_*Sin[c_.*x_^2]*FresnelC[b_.*x_],x_Symbol] :=
  -Cos[Pi/2*b^2*x^2]*FresnelC[b*x]/(Pi*b^2) + x/(2*Pi*b) +
  Dist[1/(2*Pi*b),Int[Cos[Pi*b^2*x^2],x]] /;
FreeQ[{b,c},x] && ZeroQ[c-Pi/2*b^2]
```

- **Derivation: Integration by parts**
- **Rule: If** $m \in \mathbb{Z} \land m > 1 \land \neg\,(m \bmod 4 = 0)$, **then**

$$\int x^m \operatorname{Cos}\left[\frac{\pi}{2}\ b^2 x^2\right] \operatorname{FresnelS}[b\,x]\, dx \longrightarrow \frac{x^{m-1} \operatorname{Sin}\left[\frac{\pi}{2}\ b^2 x^2\right] \operatorname{FresnelS}[b\,x]}{\pi\, b^2} - \frac{x^m}{2\, b\, m\, \pi} + \frac{1}{2\,\pi\, b} \int x^{m-1} \operatorname{Cos}\left[\pi\ b^2 x^2\right] dx - \frac{m-1}{\pi\, b^2} \int x^{m-2} \operatorname{Sin}\left[\frac{\pi}{2}\ b^2 x^2\right] \operatorname{FresnelS}[b\,x]\, dx$$

- **Program code:**

```
Int[x_^m_*Cos[c_.*x_^2]*FresnelS[b_.*x_],x_Symbol] :=
  x^(m-1)*Sin[Pi/2*b^2*x^2]*FresnelS[b*x]/(Pi*b^2) - x^m/(2*b*m*Pi) +
  Dist[1/(2*Pi*b),Int[x^(m-1)*Cos[Pi*b^2*x^2],x]] -
  Dist[(m-1)/(Pi*b^2),Int[x^(m-2)*Sin[Pi/2*b^2*x^2]*FresnelS[b*x],x]] /;
FreeQ[{b,c},x] && ZeroQ[c-Pi/2*b^2] && IntegerQ[m] && m>1 && Not[Mod[m,4]==0]
```

```
Int[x_^m_*Sin[c_.*x_^2]*FresnelC[b_.*x_],x_Symbol] :=
  -x^(m-1)*Cos[Pi/2*b^2*x^2]*FresnelC[b*x]/(Pi*b^2) + x^m/(2*b*m*Pi) +
  Dist[1/(2*Pi*b),Int[x^(m-1)*Cos[Pi*b^2*x^2],x]] +
  Dist[(m-1)/(Pi*b^2),Int[x^(m-2)*Cos[Pi/2*b^2*x^2]*FresnelC[b*x],x]] /;
FreeQ[{b,c},x] && ZeroQ[c-Pi/2*b^2] && IntegerQ[m] && m>1 && Not[Mod[m,4]==0]
```

- **Derivation: Inverted integration by parts**

- **Rule: If** $m \in \mathbb{Z} \land m < -1 \land m \bmod 4 = 2$**, then**

$$\int x^m \operatorname{Cos}\left[\frac{\pi}{2} b^2 x^2\right] \operatorname{FresnelS}[b x] \, dx \longrightarrow \frac{x^{m+1} \operatorname{Cos}\left[\frac{\pi}{2} b^2 x^2\right] \operatorname{FresnelS}[b x]}{m+1} - \frac{b}{2(m+1)} \int x^{m+1} \operatorname{Sin}\left[\pi b^2 x^2\right] dx + \frac{\pi b^2}{m+1} \int x^{m+2} \operatorname{Sin}\left[\frac{\pi}{2} b^2 x^2\right] \operatorname{FresnelS}[b x] \, dx$$

- **Program code:**

```
Int[x_^m_*Cos[c_.*x_^2]*FresnelS[b_.*x_],x_Symbol] :=
  x^(m+1)*Cos[Pi/2*b^2*x^2]*FresnelS[b*x]/(m+1) -
  Dist[b/(2*(m+1)),Int[x^(m+1)*Sin[Pi*b^2*x^2],x]] +
  Dist[Pi*b^2/(m+1),Int[x^(m+2)*Sin[Pi/2*b^2*x^2]*FresnelS[b*x],x]] /;
FreeQ[{b,c},x] && ZeroQ[c-Pi/2*b^2] && IntegerQ[m] && m<-1 && Mod[m,4]==2
```

```
Int[x_^m_*Sin[c_.*x_^2]*FresnelC[b_.*x_],x_Symbol] :=
  x^(m+1)*Sin[Pi/2*b^2*x^2]*FresnelC[b*x]/(m+1) -
  Dist[b/(2*(m+1)),Int[x^(m+1)*Sin[Pi*b^2*x^2],x]] -
  Dist[Pi*b^2/(m+1),Int[x^(m+2)*Cos[Pi/2*b^2*x^2]*FresnelC[b*x],x]] /;
FreeQ[{b,c},x] && ZeroQ[c-Pi/2*b^2] && IntegerQ[m] && m<-1 && Mod[m,4]==2
```