Integration Rules for

$\int (\sin^j(z))^m \left(A + B\sin^k(z) + C\sin^{2k}(z)\right)\left(a + b\sin^k(z)\right)^n \, dz$ when $j^2 = 1 \bigwedge k^2 = 1$

Domain Map



Legend:

- The rule number in a colored region indicates the rule to use for integrals in that region.
- The rule number next to a colored line indicates the rule to use for integrals on that line.
- A white region or line indicates there is no rule for integrals in that region or on that line.
- A solid black line indicates integrals on that line are handled by rules in another section.
- A dashed black line on the border of a region indicates integrals on that border are handled by the rule for that region.
- The arrow(s) following a rule number indicates the direction the rule drives integrands in the n×m exponent plane.
- A ⋄ following a rule number indicates the rule transforms the integrand into a form handled by another section.
- A red (stop) disk indicates the terminal rule to use for the point at the center of the disk.
- A cyan disk indicates the non-terminal rule to use for the point at the center of the disk.

Integration Rules for $\int(\sin^j(z))^m\left(A+B\sin^k(z)+C\sin^{2k}(z)\right)dz$ when $j^2=1\bigwedge k^2=1$

Rules 7 – 8: $\int(\text{Sin}[c+d\,x]^j)^m\left(A+B\,\text{Sin}[c+d\,x]^k+C\,\text{Sin}[c+d\,x]^{2k}\right)dx$

- **Derivation: Rule 7b with $B=0$ and $A+(A+C)\left(m+\frac{k+1}{2}\right)=0$**
- **Rule 7a: If $j^2=k^2=1\bigwedge A+(A+C)\left(j\,k\,m+\frac{k+1}{2}\right)=0$, then**

$$\int(\text{Sin}[c+d\,x]^j)^m\left(A+C\,\text{Sin}[c+d\,x]^{2k}\right)dx\longrightarrow\frac{A\,\text{Cos}[c+d\,x]\,(\text{Sin}[c+d\,x]^j)^{m+j\,k}}{d\left(j\,k\,m+\frac{k+1}{2}\right)}$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_*(A_+C_.*sin[c_.+d_.*x_]^k2_),x_Symbol] :=
  A*Cos[c+d*x]*(Sin[c+d*x]^j)^(m+j*k2/2)/(d*(j*k2/2*m+(k2/2+1)/2)) /;
FreeQ[{c,d,A,C,m},x] && OneQ[j^2,k2^2/4] && ZeroQ[A+(A+C)*(j*k2/2*m+(k2/2+1)/2)]
```

- **Derivation: Rule 5 with $a=1$, $b=0$ and $n=0$**
- **Rule 7b: If $j^2=k^2=1\bigwedge j\,k\,m+\frac{k+1}{2}\neq0\bigwedge j\,k\,m\leq-1$, then**

$$\int(\text{Sin}[c+d\,x]^j)^m\left(A+B\,\text{Sin}[c+d\,x]^k+C\,\text{Sin}[c+d\,x]^{2k}\right)dx\longrightarrow\frac{A\,\text{Cos}[c+d\,x]\,(\text{Sin}[c+d\,x]^j)^{m+j\,k}}{d\left(j\,k\,m+\frac{k+1}{2}\right)}+$$
$$\frac{1}{j\,k\,m+\frac{k+1}{2}}\int(\text{Sin}[c+d\,x]^j)^{m+j\,k}\left(B\left(j\,k\,m+\frac{k+1}{2}\right)+\left(A+(A+C)\left(j\,k\,m+\frac{k+1}{2}\right)\right)\text{Sin}[c+d\,x]^k\right)dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(A_.+B_.*sin[c_.+d_.*x_]^k_.+C_.*sin[c_.+d_.*x_]^k2_),x_Symbol] :=
  A*Cos[c+d*x]*(Sin[c+d*x]^j)^(m+j*k)/(d*(j*k*m+(k+1)/2)) +
  Dist[1/(j*k*m+(k+1)/2),
    Int[(sin[c+d*x]^j)^(m+j*k)*Sim[B*(j*k*m+(k+1)/2)+(A+(A+C)*(j*k*m+(k+1)/2))*sin[c+d*x]^k,x],x]] /
FreeQ[{c,d,A,B,C},x] && OneQ[j^2,k^2] && k2===2*k && RationalQ[m] && j*k*m+(k+1)/2≠0 && j*k*m≤-1
```

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(A_+C_.*sin[c_.+d_.*x_]^k2_),x_Symbol] :=
  A*Cos[c+d*x]*(Sin[c+d*x]^j)^(m+j*k2/2)/(d*(j*k2/2*m+(k2/2+1)/2)) +
   Dist[(A+(A+C)*(j*k2/2*m+(k2/2+1)/2))/(j*k2/2*m+(k2/2+1)/2),Int[(sin[c+d*x]^j)^(m+j*k2),x]] /;
FreeQ[{c,d,A,C},x] && OneQ[j^2,k2^2/4] && RationalQ[m] && j*k2/2*m+(k2/2+1)/2≠0 && j*k2/2*m≤-1
```

- **Derivation: Rule 2 with $a = 0$, $b = 1$ and $n = 0$**

- **Rule 8: If $j^2 = k^2 = 1 \bigwedge j\,k\,m + \frac{k+3}{2} \neq 0 \bigwedge j\,k\,m \geq -1$, then**

$$\int \left(\text{Sin}[c+d\,x]^j\right)^m \left(A + B\,\text{Sin}[c+d\,x]^k + C\,\text{Sin}[c+d\,x]^{2\,k}\right) dx \longrightarrow -\frac{C\,\text{Cos}[c+d\,x]\left(\text{Sin}[c+d\,x]^j\right)^{m+j\,k}}{d\left(j\,k\,m+\frac{k+3}{2}\right)} + \frac{1}{j\,k\,m+\frac{k+3}{2}} \int \left(\text{Sin}[c+d\,x]^j\right)^m \left(A + (A+C)\left(j\,k\,m+\frac{k+1}{2}\right) + B\left(j\,k\,m+\frac{k+3}{2}\right)\text{Sin}[c+d\,x]^k\right) dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(A_.+B_.*sin[c_.+d_.*x_]^k_.+C_.*sin[c_.+d_.*x_]^k2_),x_Symbol] :=
  -C*Cos[c+d*x]*(Sin[c+d*x]^j)^(m+j*k)/(d*(j*k*m+(k+3)/2)) +
  Dist[1/(j*k*m+(k+3)/2),
    Int[(sin[c+d*x]^j)^m*Sim[A+(A+C)*(j*k*m+(k+1)/2)+B*(j*k*m+(k+3)/2)*sin[c+d*x]^k,x],x]] /;
FreeQ[{c,d,A,B,C},x] && OneQ[j^2,k^2] && k2===2*k && RationalQ[m] && j*k*m+(k+3)/2≠0 && j*k*m≥-1
```

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(A_+C_.*sin[c_.+d_.*x_]^k2_),x_Symbol] :=
  -C*Cos[c+d*x]*(Sin[c+d*x]^j)^(m+j*k2/2)/(d*(j*k2/2*m+(k2/2+3)/2)) +
  Dist[(A+(A+C)*(j*k2/2*m+(k2/2+1)/2))/(j*k2/2*m+(k2/2+3)/2),Int[(sin[c+d*x]^j)^m,x]] /;
FreeQ[{c,d,A,C},x] && OneQ[j^2,k2^2/4] && RationalQ[m] && j*k2/2*m+(k2/2+3)/2≠0 && j*k2/2*m≥-1
```

Integration Rules for

$\int\left(A+B\sin^k(z)+C\sin^{2k}(z)\right)\left(a+b\sin^k(z)\right)^n\,dz$ when $k^2=1$

Rule a: $\int\frac{A+B\,\text{Sin}[c+d\,x]^k+C\,\text{Sin}[c+d\,x]^{2k}}{a+b\,\text{Sin}[c+d\,x]^k}\,dx$

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{A+B\,z+C\,z^2}{a+b\,z}=\frac{C\,z}{b}+\frac{b\,A+(b\,B-a\,C)\,z}{b\,(a+b\,z)}$
- **Rule a1:**

$$\int\frac{A+B\,\text{Sin}[c+d\,x]+C\,\text{Sin}[c+d\,x]^2}{a+b\,\text{Sin}[c+d\,x]}\,dx\longrightarrow-\frac{C\,\text{Cos}[c+d\,x]}{b\,d}+\frac{1}{b}\int\frac{b\,A+(b\,B-a\,C)\,\text{Sin}[c+d\,x]}{a+b\,\text{Sin}[c+d\,x]}\,dx$$

- **Program code:**

```
Int[(A_.+B_.*sin[c_.+d_.*x_]+C_.*sin[c_.+d_.*x_]^2)/(a_+b_.*sin[c_.+d_.*x_]),x_Symbol] :=
  -C*Cos[c+d*x]/(b*d) + Dist[1/b,Int[(b*A+(b*B-a*C)*sin[c+d*x])/(a+b*sin[c+d*x]),x]] /;
FreeQ[{a,b,c,d,A,B,C},x]
```

```
Int[(A_+C_.*sin[c_.+d_.*x_]^2)/(a_+b_.*sin[c_.+d_.*x_]),x_Symbol] :=
  -C*Cos[c+d*x]/(b*d) + Dist[1/b,Int[(b*A-a*C*sin[c+d*x])/(a+b*sin[c+d*x]),x]] /;
FreeQ[{a,b,c,d,A,C},x]
```

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{A+B\,z^{-1}+C\,z^{-2}}{a+b\,z^{-1}}=\frac{A}{a}+\frac{a\,C-(b\,A-a\,B)\,z}{a\,z\,(b+a\,z)}$
- **Rule a2: If $a^2-b^2\neq0$, then**

$$\int\frac{A+B\,\text{Csc}[c+d\,x]+C\,\text{Csc}[c+d\,x]^2}{a+b\,\text{Csc}[c+d\,x]}\,dx\longrightarrow\frac{A\,x}{a}+\int\frac{C+(B-b\,A/a)\,\text{Sin}[c+d\,x]}{\text{Sin}[c+d\,x]\,(b+a\,\text{Sin}[c+d\,x])}\,dx$$

- **Program code:**

```
Int[(A_.+B_.*sin[c_.+d_.*x_]^(-1)+C_.*sin[c_.+d_.*x_]^(-2))/(a_+b_.*sin[c_.+d_.*x_]^(-1)),x_Symbol]
  A*x/a + Int[(C+(B-b*A/a)*sin[c+d*x])/(sin[c+d*x]*(b+a*sin[c+d*x])),x] /;
FreeQ[{a,b,c,d,A,B,C},x] && NonzeroQ[a^2-b^2]
```

```
Int[(A_+C_.*sin[c_.+d_.*x_]^(-2))/(a_+b_.*sin[c_.+d_.*x_]^(-1)),x_Symbol] :=
  A*x/a + Int[(C-b*A/a*sin[c+d*x])/(sin[c+d*x]*(b+a*sin[c+d*x])),x] /;
FreeQ[{a,b,c,d,A,C},x] && NonzeroQ[a^2-b^2]
```

$$\int \left(A + B \operatorname{Csc}[c + d x] + C \operatorname{Csc}[c + d x]^2\right) (a + b \operatorname{Csc}[c + d x])^{n/2} \, dx$$

- **Derivation: Piecewise constant extraction**
- **Basis:** $\partial_z \left(\sqrt{z}\ \sqrt{1/z}\right) = 0$
- **Rule: If** $n^2 = 1$, **then**

$$\int \left(A + B \operatorname{Csc}[c + d x] + C \operatorname{Csc}[c + d x]^2\right) (b \operatorname{Csc}[c + d x])^{n/2} \, dx \longrightarrow$$

$$(\operatorname{Sin}[c + d x])^{n/2} (b \operatorname{Csc}[c + d x])^{n/2} \int \frac{C + B \operatorname{Sin}[c + d x] + A \operatorname{Sin}[c + d x]^2}{\operatorname{Sin}[c + d x]^{n/2+2}} \, dx$$

- **Program code:**

```
Int[(A_+B_.*sin[c_.+d_.*x_]^(-1)+C_.*sin[c_.+d_.*x_]^(-2))*(b_.*sin[c_.+d_.*x_]^(-1))^n_,x_Symbol] :
  Dist[Sin[c+d*x]^n*(b*Csc[c+d*x])^n,Int[(C+B*sin[c+d*x]+A*sin[c+d*x]^2)/sin[c+d*x]^(n+2),x]] /;
FreeQ[{b,c,d,A,B,C},x] && ZeroQ[n^2-1/4]
```

```
Int[(A_+C_.*sin[c_.+d_.*x_]^(-2))*(b_.*sin[c_.+d_.*x_]^(-1))^n_,x_Symbol] :=
  Dist[Sin[c+d*x]^n*(b*Csc[c+d*x])^n,Int[(C+A*sin[c+d*x]^2)/sin[c+d*x]^(n+2),x]] /;
FreeQ[{b,c,d,A,C},x] && ZeroQ[n^2-1/4]
```

- **Derivation: Piecewise constant extraction**
- **Basis:** $\partial_z \frac{\sqrt{b+a f[z]}}{\sqrt{f[z]}\ \sqrt{a+b/f[z]}} = 0$
- **Rule: If** $a^2 - b^2 \neq 0 \bigwedge n - \frac{1}{2} \in \mathbb{Z} \bigwedge -2 < n < 0$, **then**

$$\int \left(A + B \operatorname{Csc}[c + d x] + C \operatorname{Csc}[c + d x]^2\right) (a + b \operatorname{Csc}[c + d x])^n \, dx \longrightarrow$$

$$\frac{\sqrt{b + a \operatorname{Sin}[c + d x]}}{\sqrt{\operatorname{Sin}[c + d x]}\ \sqrt{a + b \operatorname{Csc}[c + d x]}} \int \frac{\left(C + B \operatorname{Sin}[c + d x] + A \operatorname{Sin}[c + d x]^2\right) (b + a \operatorname{Sin}[c + d x])^n}{\operatorname{Sin}[c + d x]^{n+2}} \, dx$$

- **Program code:**

```
Int[(A_.+B_.*sin[c_.+d_.*x_]^(-1)+C_.*sin[c_+d_.*x_]^(-2))*(a_.+b_.*sin[c_.+d_.*x_]^(-1))^n_,x_Symbo
  Dist[Sqrt[b+a*Sin[c+d*x]]/(Sqrt[Sin[c+d*x]]*Sqrt[a+b*Csc[c+d*x]]),
    Int[(C+B*sin[c+d*x]+A*sin[c+d*x]^2)*(b+a*sin[c+d*x])^n/sin[c+d*x]^(n+2),x]] /;
FreeQ[{a,b,c,d,A,B,C},x] && NonzeroQ[a^2-b^2] && IntegerQ[n-1/2] && -2<n<0
```

```
Int[(A_.+C_.*sin[c_.+d_.*x_]^(-2))*(a_+b_.*sin[c_.+d_.*x_]^(-1))^n_,x_Symbol] :=
  Dist[Sqrt[b+a*Sin[c+d*x]]/(Sqrt[Sin[c+d*x]]*Sqrt[a+b*Csc[c+d*x]]),
    Int[(C+A*sin[c+d*x]^2)*(b+a*sin[c+d*x])^n/sin[c+d*x]^(n+2),x]] /;
FreeQ[{a,b,c,d,A,C},x] && NonzeroQ[a^2-b^2] && IntegerQ[n-1/2] && -2<n<0
```

$$\int \left(A + B\,\mathrm{Sin}[c + d\,x]^k + C\,\mathrm{Sin}[c + d\,x]^{2\,k}\right)\,\left(a + b\,\mathrm{Sin}[c + d\,x]^k\right)^n\,dx$$

- **Derivation: Algebraic simplification**
- **Basis: If** $a^2\,C - a\,b\,B + b^2\,A = 0$, **then** $A + B\,z + C\,z^2 = \frac{1}{b^2}\,(b\,B - a\,C + b\,C\,z)\,(a + b\,z)$
- **Rule: If** $k^2 = 1\,\wedge\,a^2\,C - a\,b\,B + b^2\,A = 0\,\wedge\,n < -1$, **then**

$$\int \left(A + B\,\mathrm{Sin}[c + d\,x]^k + C\,\mathrm{Sin}[c + d\,x]^{2\,k}\right)\,\left(a + b\,\mathrm{Sin}[c + d\,x]^k\right)^n\,dx \longrightarrow$$
$$\frac{1}{b^2}\int \left(b\,B - a\,C + b\,C\,\mathrm{Sin}[c + d\,x]^k\right)\,\left(a + b\,\mathrm{Sin}[c + d\,x]^k\right)^{n+1}\,dx$$

- **Program code:**

```
Int[(A_.+B_.*sin[c_.+d_.*x_]^k_.+C_.*sin[c_.+d_.*x_]^k2_)*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  Dist[1/b^2,Int[Sim[b*B-a*C+b*C*sin[c+d*x]^k,x]*(a+b*sin[c+d*x]^k)^(n+1),x]] /;
FreeQ[{a,b,c,d,A,B,C},x] && OneQ[k^2] && k2===2*k && ZeroQ[a^2*C-a*b*B+b^2*A] && RationalQ[n] &&
  n<-1
```

```
Int[(A_+C_.*sin[c_.+d_.*x_]^k2_)*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_.,x_Symbol] :=
  Dist[C/b^2,Int[Sim[-a+b*sin[c+d*x]^k,x]*(a+b*sin[c+d*x]^k)^(n+1),x]] /;
FreeQ[{a,b,c,d,A,C},x] && OneQ[k^2] && k2===2*k && ZeroQ[a^2*C+b^2*A] && RationalQ[n] && n<-1
```

Rules 17 – 18: $\int \left(A + B\,\mathrm{Csc}[c + d\,x] + C\,\mathrm{Csc}[c + d\,x]^2\right) (a + b\,\mathrm{Csc}[c + d\,x])^n\,dx$

- **Derivation: Rule 6 with** $m = 0$ **and** $k = -1$

- **Rule 17: If** $a^2 - b^2 \neq 0 \land a^2 C - a\,b\,B + b^2 A \neq 0 \land n < -1$, **then**

$$\int \left(A + B\,\mathrm{Csc}[c + d\,x] + C\,\mathrm{Csc}[c + d\,x]^2\right) (a + b\,\mathrm{Csc}[c + d\,x])^n\,dx \longrightarrow$$

$$\frac{\left(a^2 C - a\,b\,B + b^2 A\right) \mathrm{Cot}[c + d\,x]\,(a + b\,\mathrm{Csc}[c + d\,x])^{n+1}}{a\,d\,(n + 1)\,\left(a^2 - b^2\right)} + \frac{1}{a\,(n + 1)\,\left(a^2 - b^2\right)} \cdot$$

$$\int \left(A\,\left(a^2 - b^2\right) (n + 1) - a\,(b\,A - a\,B + b\,C)\,(n + 1)\,\mathrm{Csc}[c + d\,x] + \left(a^2 C - a\,b\,B + b^2 A\right) (n + 2)\,\mathrm{Csc}[c + d\,x]^2\right)$$
$$(a + b\,\mathrm{Csc}[c + d\,x])^{n+1}\,dx$$

- **Program code:**

```
Int[(A_.+B_.*sin[c_.+d_.*x_]^(-1)+C_.*sin[c_.+d_.*x_]^(-2))*(a_+b_.*sin[c_.+d_.*x_]^(-1))^n_,x_Symbo
  (a^2*C-a*b*B+b^2*A)*Cot[c+d*x]*(a+b*Csc[c+d*x])^(n+1)/(a*d*(n+1)*(a^2-b^2)) +
  Dist[1/(a*(n+1)*(a^2-b^2)),
    Int[Sim[A*(a^2-b^2)*(n+1)-(a*(b*A-a*B+b*C)*(n+1))*sin[c+d*x]^(-1)+
        (a^2*C-a*b*B+b^2*A)*(n+2)*sin[c+d*x]^(-2),x]*
      (a+b*sin[c+d*x]^(-1))^(n+1),x]] /;
FreeQ[{a,b,c,d,A,B,C},x] && NonzeroQ[a^2-b^2] && NonzeroQ[a^2*C-a*b*B+b^2*A] && RationalQ[n] && n<-1
```

```
Int[(A_.+C_.*sin[c_.+d_.*x_]^(-2))*(a_+b_.*sin[c_.+d_.*x_]^(-1))^n_,x_Symbol] :=
  (a^2*C+b^2*A)*Cot[c+d*x]*(a+b*Csc[c+d*x])^(n+1)/(a*d*(n+1)*(a^2-b^2)) +
  Dist[1/(a*(n+1)*(a^2-b^2)),
    Int[Sim[A*(a^2-b^2)*(n+1)-(a*b*(A+C)*(n+1))*sin[c+d*x]^(-1)+(a^2*C+b^2*A)*(n+2)*sin[c+d*x]^(-2),
      (a+b*sin[c+d*x]^(-1))^(n+1),x]] /;
FreeQ[{a,b,c,d,A,C},x] && NonzeroQ[a^2-b^2] && NonzeroQ[a^2*C+b^2*A] && RationalQ[n] && n<-1
```

- **Derivation: Rule 3 with $m = 0$ and $k = -1$**
- **Note: If $A = B = a^2 - b^2 = 0$, there is an $a^2 - b^2 = 0$ rule that simplifies resulting integrand to $(a + b\,\text{Csc}[c + d\,x])^n$.**
- **Rule 18: If $n > 0 \bigwedge \neg\,(A = B = a^2 - b^2 = 0)$, then**

$$\int \left(A + B\,\text{Csc}[c + d\,x] + C\,\text{Csc}[c + d\,x]^2\right)\,(a + b\,\text{Csc}[c + d\,x])^n\,dx \longrightarrow$$
$$-\frac{C\,\text{Cot}[c + d\,x]\,(a + b\,\text{Csc}[c + d\,x])^n}{d\,(n + 1)} + \frac{1}{n + 1}\,\cdot$$
$$\int \left(a\,A\,(n + 1) + (b\,A + a\,B + n\,(b\,A + a\,B + b\,C))\,\text{Csc}[c + d\,x] + (a\,C\,n + b\,B\,(n + 1))\,\text{Csc}[c + d\,x]^2\right)$$
$$(a + b\,\text{Csc}[c + d\,x])^{n-1}\,dx$$

- **Program code:**

```
Int[(A_.+B_.*sin[c_.+d_.*x_]^(-1)+C_.*sin[c_.+d_.*x_]^(-2))*(a_+b_.*sin[c_.+d_.*x_]^(-1))^n_.,x_Symb
  -C*Cot[c+d*x]*(a+b*Csc[c+d*x])^n/(d*(n+1)) +
  Dist[1/(n+1),
    Int[Sim[a*A*(n+1)+(b*A+a*B+n*(b*A+a*B+b*C))*sin[c+d*x]^(-1)+(a*C*n+b*B*(n+1))*sin[c+d*x]^(-2),x]
      (a+b*sin[c+d*x]^(-1))^(n-1),x]] /;
FreeQ[{a,b,c,d,A,B,C},x] && RationalQ[n] && n>0
```

```
Int[(A_.+C_.*sin[c_.+d_.*x_]^(-2))*(a_+b_.*sin[c_.+d_.*x_]^(-1))^n_.,x_Symbol] :=
  -C*Cot[c+d*x]*(a+b*Csc[c+d*x])^n/(d*(n+1)) +
  Dist[1/(n+1),
    Int[Sim[a*A*(n+1)+b*(A+n*(A+C))*sin[c+d*x]^(-1)+a*C*n*sin[c+d*x]^(-2),x]*
      (a+b*sin[c+d*x]^(-1))^(n-1),x]] /;
FreeQ[{a,b,c,d,A,C},x] && RationalQ[n] && n>0 && Not[ZeroQ[A] && ZeroQ[a^2-b^2]]
```

Rules 15 – 16: $\int \left(A + B\,\text{Sin}[c + d\,x]^k + C\,\text{Sin}[c + d\,x]^{2\,k}\right) \left(b\,\text{Sin}[c + d\,x]^k\right)^n \, dx$

- **Derivation: For $k = 1$, Rule 9b with $k = 1$**
- **Derivation: For $k = -1$, ???**
- **Rule 15: If $k^2 = 1 \wedge n < -1$, then**

$$\int \left(A + B\,\text{Sin}[c + d\,x]^k + C\,\text{Sin}[c + d\,x]^{2\,k}\right) \left(b\,\text{Sin}[c + d\,x]^k\right)^n dx \longrightarrow \frac{2\,A\,\text{Cos}[c + d\,x] \left(b\,\text{Sin}[c + d\,x]^k\right)^{n+1}}{b\,d\,(2\,n + k + 1)} + \frac{1}{b\,(2\,n + k + 1)} \int \left((2\,n + k + 1)\,B + (2\,A + (A + C)\,(2\,n + k + 1))\,\text{Sin}[c + d\,x]^k\right) \left(b\,\text{Sin}[c + d\,x]^k\right)^{n+1} dx$$

- **Program code:**

```
Int[(A_.+B_.*sin[c_.+d_.*x_]^k_.+C_.*sin[c_.+d_.*x_]^k2_)*(b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  2*A*Cos[c+d*x]*(b*Sin[c+d*x]^k)^(n+1)/(b*d*(2*n+k+1)) +
  Dist[1/(b*(2*n+k+1)),
    Int[Sim[(2*n+k+1)*B+(2*A+(A+C)*(2*n+k+1))*sin[c+d*x]^k,x]*(b*sin[c+d*x]^k)^(n+1),x]] /;
FreeQ[{b,c,d,A,B,C},x] && OneQ[k^2] && k2===2*k  && RationalQ[n] && n<-1
```

```
Int[(A_+C_.*sin[c_.+d_.*x_]^k2_)*(b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  2*A*Cos[c+d*x]*(b*Sin[c+d*x]^k)^(n+1)/(b*d*(2*n+k+1)) +
  Dist[(2*A+(A+C)*(2*n+k+1))/(b^2*(2*n+k+1)),Int[(b*sin[c+d*x]^k)^(n+2),x]] /;
FreeQ[{b,c,d,A,C},x] && OneQ[k^2] && k2===2*k  && RationalQ[n] && n<-1
```

- **Derivation: Rule 2 or 3 with $m = 0$ and $a = 0$**
- **Rule 16: If $k^2 = 1 \wedge n > -1$, then**

$$\int \left(A + B\,\text{Sin}[c + d\,x]^k + C\,\text{Sin}[c + d\,x]^{2\,k}\right) \left(b\,\text{Sin}[c + d\,x]^k\right)^n dx \longrightarrow -\frac{2\,C\,\text{Cos}[c + d\,x] \left(b\,\text{Sin}[c + d\,x]^k\right)^{n+1}}{b\,d\,(2\,n + k + 3)} + \frac{1}{2\,n + k + 3} \int \left(2\,A + (A + C)\,(2\,n + k + 1) + B\,(2\,n + k + 3)\,\text{Sin}[c + d\,x]^k\right) \left(b\,\text{Sin}[c + d\,x]^k\right)^n dx$$

- **Program code:**

```
Int[(A_.+B_.*sin[c_.+d_.*x_]^k_.+C_.*sin[c_.+d_.*x_]^k2_)*(b_.*sin[c_.+d_.*x_]^k_.)^n_.,x_Symbol] :=
  -2*C*Cos[c+d*x]*(b*Sin[c+d*x]^k)^(n+1)/(b*d*(2*n+k+3)) +
  Dist[1/(2*n+k+3),
    Int[Sim[2*A+(A+C)*(2*n+k+1)+B*(2*n+k+3)*sin[c+d*x]^k,x]*(b*sin[c+d*x]^k)^n,x]] /;
FreeQ[{b,c,d,A,B,C},x] && OneQ[k^2] && k2===2*k && RationalQ[n] && n>-1
```

```
Int[(A_+C_.*sin[c_.+d_.*x_]^k2_)*(b_.*sin[c_.+d_.*x_]^k_.)^n_.,x_Symbol] :=
  -2*C*Cos[c+d*x]*(b*Sin[c+d*x]^k)^(n+1)/(b*d*(2*n+k+3)) +
  Dist[(2*A+(A+C)*(2*n+k+1))/(2*n+k+3),Int[(b*sin[c+d*x]^k)^n,x]] /;
FreeQ[{b,c,d,A,C},x] && OneQ[k^2] && k2===2*k && RationalQ[n] && n>-1
```

Integration Rules for

$$\int\left(\sin^j(z)\right)^m\left(A+B\sin^k(z)+C\sin^{2k}(z)\right)\left(a+b\sin^k(z)\right)^n\,dz \text{ when } j^2=1\bigwedge k^2=1$$

Rule b: $\int\frac{A+B\,\text{Sin}[c+d\,x]+C\,\text{Sin}[c+d\,x]^2}{\text{Sin}[c+d\,x]\,(a+b\,\text{Sin}[c+d\,x])}\,dx$

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{A+B\,z+C\,z^2}{z\,(a+b\,z)}=\frac{A}{a\,z}-\frac{b\,A-a\,B-a\,C\,z}{a\,(a+b\,z)}$
- **Rule b: If** $a^2-b^2\neq 0$**, then**

$$\int\frac{A+B\,\text{Sin}[c+d\,x]+C\,\text{Sin}[c+d\,x]^2}{\text{Sin}[c+d\,x]\,(a+b\,\text{Sin}[c+d\,x])}\,dx\;\longrightarrow\;\frac{A}{a}\int\frac{1}{\text{Sin}[c+d\,x]}\,dx-\frac{1}{a}\int\frac{b\,A-a\,B-a\,C\,\text{Sin}[c+d\,x]}{a+b\,\text{Sin}[c+d\,x]}\,dx$$

- **Program code:**

```
Int[(A_.+B_.*sin[c_.+d_.*x_]+C_.*sin[c_.+d_.*x_]^2)/(sin[c_.+d_.*x_]*(a_+b_.*sin[c_.+d_.*x_])),x_Sym
  A/a*Int[1/sin[c+d*x],x] -
  Dist[1/a,Int[(b*A-a*B-a*C*sin[c+d*x])/(a+b*sin[c+d*x]),x]] /;
FreeQ[{a,b,c,d,A,B,C},x] && NonzeroQ[a^2-b^2]
```

```
Int[(A_+C_.*sin[c_.+d_.*x_]^2)/(sin[c_.+d_.*x_]*(a_+b_.*sin[c_.+d_.*x_])),x_Symbol] :=
  A/a*Int[1/sin[c+d*x],x] -
  Dist[1/a,Int[(b*A-a*C*sin[c+d*x])/(a+b*sin[c+d*x]),x]] /;
FreeQ[{a,b,c,d,A,C},x] && NonzeroQ[a^2-b^2]
```

Rule c: $\int \frac{A + B \operatorname{Sin}[c + d x] + C \operatorname{Sin}[c + d x]^2}{\sqrt{\operatorname{Sin}[c + d x]} \, (a + b \operatorname{Sin}[c + d x])} \, dx$

- **Derivation: Algebraic expansion**
- **Rule c: If $a^2 - b^2 \neq 0$, then**

$$\int \frac{A + B \operatorname{Sin}[c + d x] + C \operatorname{Sin}[c + d x]^2}{\sqrt{\operatorname{Sin}[c + d x]} \, (a + b \operatorname{Sin}[c + d x])} \, dx \longrightarrow$$

$$\frac{C}{b} \int \sqrt{\operatorname{Sin}[c + d x]} \, dx + \frac{1}{b} \int \frac{b A + (b B - a C) \operatorname{Sin}[c + d x]}{\sqrt{\operatorname{Sin}[c + d x]} \, (a + b \operatorname{Sin}[c + d x])} \, dx$$

- **Program code:**

```
Int[(A_.+B_.*sin[c_.+d_.*x_]+C_.*sin[c_.+d_.*x_]^2)/(Sqrt[sin[c_.+d_.*x_]]*(a_+b_.*sin[c_.+d_.*x_])),x_Symbol] :=
  Dist[C/b,Int[Sqrt[sin[c+d*x]],x]] +
  Dist[1/b,Int[Sim[b*A+(b*B-a*C)*sin[c+d*x],x]/(Sqrt[sin[c+d*x]]*(a+b*sin[c+d*x])),x]] /;
FreeQ[{a,b,c,d,A,B,C},x] && NonzeroQ[a^2-b^2]
```

```
Int[(A_+C_.*sin[c_.+d_.*x_]^2)/(Sqrt[sin[c_.+d_.*x_]]*(a_+b_.*sin[c_.+d_.*x_])),x_Symbol] :=
  Dist[C/b,Int[Sqrt[sin[c+d*x]],x]] +
  Dist[1/b,Int[(b*A-a*C*sin[c+d*x])/(Sqrt[sin[c+d*x]]*(a+b*sin[c+d*x])),x]] /;
FreeQ[{a,b,c,d,A,C},x] && NonzeroQ[a^2-b^2]
```

Rule d: $\int \frac{A + B \operatorname{Sin}[c + d x] + C \operatorname{Sin}[c + d x]^2}{\operatorname{Sin}[c + d x] \sqrt{a + b \operatorname{Sin}[c + d x]}} \, dx$

- **Derivation: Algebraic expansion**
- **Rule d: If $a^2 - b^2 \neq 0$, then**

$$\int \frac{A + B \operatorname{Sin}[c + d x] + C \operatorname{Sin}[c + d x]^2}{\operatorname{Sin}[c + d x] \sqrt{a + b \operatorname{Sin}[c + d x]}} \, dx \longrightarrow$$

$$\frac{C}{b} \int \sqrt{a + b \operatorname{Sin}[c + d x]} \, dx + \frac{1}{b} \int \frac{b A + (b B - a C) \operatorname{Sin}[c + d x]}{\operatorname{Sin}[c + d x] \sqrt{a + b \operatorname{Sin}[c + d x]}} \, dx$$

- **Program code:**

```
Int[(A_.+B_.*sin[c_.+d_.*x_]+C_.*sin[c_.+d_.*x_]^2)/(sin[c_.+d_.*x_]*Sqrt[a_+b_.*sin[c_.+d_.*x_]]),x
  Dist[C/b,Int[Sqrt[a+b*sin[c+d*x]],x]] +
  Dist[1/b,Int[Sim[b*A+(b*B-a*C)*sin[c+d*x],x]/(sin[c+d*x]*Sqrt[a+b*sin[c+d*x]]),x]] /;
FreeQ[{a,b,c,d,A,B,C},x] && NonzeroQ[a^2-b^2]
```

```
Int[(A_+C_.*sin[c_.+d_.*x_]^2)/(sin[c_.+d_.*x_]*Sqrt[a_+b_.*sin[c_.+d_.*x_]]),x_Symbol] :=
  Dist[C/b,Int[Sqrt[a+b*sin[c+d*x]],x]] +
  Dist[1/b,Int[(A*b-a*C*sin[c+d*x])/(sin[c+d*x]*Sqrt[a+b*sin[c+d*x]]),x]] /;
FreeQ[{a,b,c,d,A,C},x] && NonzeroQ[a^2-b^2]
```

$$\text{Rule e:}\quad \int \frac{A + B \,\text{Sin}[c + d\,x] + C \,\text{Sin}[c + d\,x]^2}{\sqrt{\text{Sin}[c + d\,x]}\,\sqrt{a + b \,\text{Sin}[c + d\,x]}}\,dx$$

- **Rule e1: If** $a^2 - b^2 \neq 0 \land 2\,b\,A - a\,C = 0$**, then**

$$\int \frac{A + A \,\text{Sin}[c + d\,x] + C \,\text{Sin}[c + d\,x]^2}{\sqrt{\text{Sin}[c + d\,x]}\,\sqrt{a + b \,\text{Sin}[c + d\,x]}}\,dx \longrightarrow$$

$$\frac{C \sqrt{\text{Sin}[c + d\,x]}\,\sqrt{a + b \,\text{Sin}[c + d\,x]}\,\text{Tan}\left[\frac{1}{2}\left(c - \frac{\pi}{2} + d\,x\right)\right]}{b\,d} + \frac{C}{b}\int \frac{\sqrt{a + b \,\text{Sin}[c + d\,x]}}{\sqrt{\text{Sin}[c + d\,x]}\,(1 + \text{Sin}[c + d\,x])}\,dx$$

- **Program code:**

```
Int[(A_+B_.*sin[c_.+d_.*x_]+C_.*sin[c_.+d_.*x_]^2)/(Sqrt[sin[c_.+d_.*x_]]*Sqrt[a_+b_.*sin[c_.+d_.*x_
  C*Sqrt[Sin[c+d*x]]*Sqrt[a+b*Sin[c+d*x]]*Tan[(c-Pi/2+d*x)/2]/(b*d) +
  C/b*Int[Sqrt[a+b*sin[c+d*x]]/(Sqrt[sin[c+d*x]]*(1+sin[c+d*x])),x] /;
FreeQ[{a,b,c,d,A,B,C},x] && NonzeroQ[a^2-b^2] && ZeroQ[A-B] && ZeroQ[2*b*A-a*C]
```

- **Derivation: Algebraic expansion**

- **Rule e2: If** $a^2 - b^2 \neq 0$**, then**

$$\int \frac{A + C \,\text{Sin}[c + d\,x]^2}{\sqrt{\text{Sin}[c + d\,x]}\,\sqrt{a + b \,\text{Sin}[c + d\,x]}}\,dx \longrightarrow$$

$$A \int \frac{1}{\sqrt{\text{Sin}[c + d\,x]}\,\sqrt{a + b \,\text{Sin}[c + d\,x]}}\,dx + C \int \frac{\text{Sin}[c + d\,x]^{3/2}}{\sqrt{a + b \,\text{Sin}[c + d\,x]}}\,dx$$

- **Program code:**

```
Int[(A_+C_.*sin[c_.+d_.*x_]^2)/(Sqrt[sin[c_.+d_.*x_]]*Sqrt[a_+b_.*sin[c_.+d_.*x_]]),x_Symbol] :=
  A*Int[1/(Sqrt[sin[c+d*x]]*Sqrt[a+b*sin[c+d*x]]),x] +
  Dist[C,Int[sin[c+d*x]^(3/2)/Sqrt[a+b*sin[c+d*x]],x]] /;
FreeQ[{a,b,c,d,A,C},x] && NonzeroQ[a^2-b^2]
```

- **Derivation: Algebraic expansion**
- **Rule e3: If $a^2 - b^2 \neq 0$, then**

$$\int \frac{A + B\,\text{Sin}[c + d\,x] + C\,\text{Sin}[c + d\,x]^2}{\sqrt{\text{Sin}[c + d\,x]}\,\sqrt{a + b\,\text{Sin}[c + d\,x]}}\,dx \longrightarrow$$

$$\frac{1}{2\,b}\int \frac{2\,b\,A - a\,C + (2\,b\,B - a\,C)\,\text{Sin}[c + d\,x]}{\sqrt{\text{Sin}[c + d\,x]}\,\sqrt{a + b\,\text{Sin}[c + d\,x]}}\,dx + \frac{C}{2\,b}\int \frac{a + a\,\text{Sin}[c + d\,x] + 2\,b\,\text{Sin}[c + d\,x]^2}{\sqrt{\text{Sin}[c + d\,x]}\,\sqrt{a + b\,\text{Sin}[c + d\,x]}}\,dx$$

- **Program code:**

```
Int[(A_.+B_.*sin[c_.+d_.*x_]+C_.*sin[c_.+d_.*x_]^2)/(Sqrt[sin[c_.+d_.*x_]]*Sqrt[a_+b_.*sin[c_.+d_.*x
  Dist[1/(2*b),Int[(2*b*A-a*C+(2*b*B-a*C)*sin[c+d*x])/(Sqrt[sin[c+d*x]]*Sqrt[a+b*sin[c+d*x]]),x]] +
  Dist[C/(2*b),Int[(a+a*sin[c+d*x]+2*b*sin[c+d*x]^2)/(Sqrt[sin[c+d*x]]*Sqrt[a+b*sin[c+d*x]]),x]] /;
FreeQ[{a,b,c,d,A,B,C},x] && NonzeroQ[a^2-b^2]
```

Rule f: $\int \frac{A + B \operatorname{Sin}[c + d x] + C \operatorname{Sin}[c + d x]^2}{\operatorname{Sin}[c + d x]^{3/2} \sqrt{a + b \operatorname{Sin}[c + d x]}} \, dx$

- **Derivation: Algebraic expansion**
- **Note: This rule is not essential, but produces simpler results.**
- **Rule f: If $a^2 - b^2 \neq 0$, then**

$$\int \frac{A + B \operatorname{Sin}[c + d x] + C \operatorname{Sin}[c + d x]^2}{\operatorname{Sin}[c + d x]^{3/2} \sqrt{a + b \operatorname{Sin}[c + d x]}} \, dx \longrightarrow$$
$$C \int \frac{1 + \operatorname{Sin}[c + d x]}{\sqrt{\operatorname{Sin}[c + d x]} \sqrt{a + b \operatorname{Sin}[c + d x]}} \, dx + \int \frac{A + (B - C) \operatorname{Sin}[c + d x]}{\operatorname{Sin}[c + d x]^{3/2} \sqrt{a + b \operatorname{Sin}[c + d x]}} \, dx$$

- **Program code:**

```
Int[(A_.+B_.*sin[c_.+d_.*x_]+C_.*sin[c_.+d_.*x_]^2)/(sin[c_.+d_.*x_]^(3/2)*Sqrt[a_+b_.*sin[c_.+d_.*x
  Dist[C,Int[(1+sin[c+d*x])/(Sqrt[sin[c+d*x]]*Sqrt[a+b*sin[c+d*x]]),x]] +
  Int[(A+(B-C)*sin[c+d*x])/(sin[c+d*x]^(3/2)*Sqrt[a+b*sin[c+d*x]]),x] /;
FreeQ[{a,b,c,d,A,B,C},x] && NonzeroQ[a^2-b^2]
```

```
Int[(A_+C_.*sin[c_.+d_.*x_]^2)/(sin[c_.+d_.*x_]^(3/2)*Sqrt[a_+b_.*sin[c_.+d_.*x_]]),x_Symbol] :=
  Dist[C,Int[(1+sin[c+d*x])/(Sqrt[sin[c+d*x]]*Sqrt[a+b*sin[c+d*x]]),x]] +
  Int[(A-C*sin[c+d*x])/(sin[c+d*x]^(3/2)*Sqrt[a+b*sin[c+d*x]]),x] /;
FreeQ[{a,b,c,d,A,C},x] && NonzeroQ[a^2-b^2]
```

Rule g: $\int \frac{A + B \sin[c + d x] + C \sin[c + d x]^2}{\sqrt{\sin[c + d x]} (a + b \sin[c + d x])^{3/2}} \, dx$

- **Derivation: Algebraic expansion**

- **Note: This rule is not essential, but produces simpler results.**

- **Rule g: If $a^2 - b^2 \neq 0$, then**

$$\int \frac{A + B \sin[c + d x] + C \sin[c + d x]^2}{\sqrt{\sin[c + d x]} (a + b \sin[c + d x])^{3/2}} \, dx \longrightarrow$$
$$\frac{C}{b} \int \frac{1 + \sin[c + d x]}{\sqrt{\sin[c + d x]} \sqrt{a + b \sin[c + d x]}} \, dx + \frac{1}{b} \int \frac{b A - a C + (b B - C (a + b)) \sin[c + d x]}{\sqrt{\sin[c + d x]} (a + b \sin[c + d x])^{3/2}} \, dx$$

- **Program code:**

```
Int[(A_.+B_.*sin[c_.+d_.*x_]+C_.*sin[c_.+d_.*x_]^2)/(Sqrt[sin[c_.+d_.*x_]]*(a_+b_.*sin[c_.+d_.*x_])^
  Dist[C/b,Int[(1+sin[c+d*x])/(Sqrt[sin[c+d*x]]*Sqrt[a+b*sin[c+d*x]]),x]] +
  Dist[1/b,Int[(b*A-a*C+(b*B-C*(a+b))*sin[c+d*x])/(Sqrt[sin[c+d*x]]*(a+b*sin[c+d*x])^(3/2)),x]] /;
FreeQ[{a,b,c,d,A,B,C},x] && NonzeroQ[a^2-b^2]
```

```
Int[(A_+C_.*sin[c_.+d_.*x_]^2)/(Sqrt[sin[c_.+d_.*x_]]*(a_+b_.*sin[c_.+d_.*x_])^(3/2)),x_Symbol] :=
  Dist[C/b,Int[(1+sin[c+d*x])/(Sqrt[sin[c+d*x]]*Sqrt[a+b*sin[c+d*x]]),x]] +
  Dist[1/b,Int[(b*A-a*C-C*(a+b)*sin[c+d*x])/(Sqrt[sin[c+d*x]]*(a+b*sin[c+d*x])^(3/2)),x]] /;
FreeQ[{a,b,c,d,A,C},x] && NonzeroQ[a^2-b^2]
```

Rule h:

$$\int \left(\text{Sin}[c+d\,x]^j\right)^m \left(A+B\,\text{Sin}[c+d\,x]^k+C\,\text{Sin}[c+d\,x]^{2\,k}\right) \left(b\,\text{Sin}[c+d\,x]^k\right)^n \,dx$$

- **Derivation: Algebraic simplification**
- **Rule h1: If** $k^2 = 1 \bigwedge m \in \mathbb{Z}$**, then**

$$\int \text{Sin}[c+d\,x]^m \left(A+B\,\text{Sin}[c+d\,x]^k+C\,\text{Sin}[c+d\,x]^{2\,k}\right) \left(b\,\text{Sin}[c+d\,x]^k\right)^n dx \longrightarrow$$
$$\frac{1}{b^{k\,m}} \int \left(A+B\,\text{Sin}[c+d\,x]^k+C\,\text{Sin}[c+d\,x]^{2\,k}\right) \left(b\,\text{Sin}[c+d\,x]^k\right)^{k\,m+n} dx$$

- **Program code:**

```
Int[sin[c_.+d_.*x_]^m_.*(A_.+B_.*sin[c_.+d_.*x_]^k_.+C_.*sin[c_.+d_.*x_]^k2_)*
    (b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  Dist[1/b^(k*m),Int[(A+B*sin[c+d*x]^k+C*sin[c+d*x]^(2*k))*(b*sin[c+d*x]^k)^(k*m+n),x]] /;
FreeQ[{b,c,d,A,B,C,n},x] && OneQ[k^2] && k2===2*k && IntegerQ[m]
```

```
Int[sin[c_.+d_.*x_]^m_.*(A_+C_.*sin[c_.+d_.*x_]^k2_)*(b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  Dist[1/b^(k*m),Int[(A+C*sin[c+d*x]^(2*k))*(b*sin[c+d*x]^k)^(k*m+n),x]] /;
FreeQ[{b,c,d,A,C,n},x] && OneQ[k^2] && k2===2*k && IntegerQ[m]
```

- **Derivation: Piecewise constant extraction**
- **Basis: If** $j^2 = 1$**, then** $\partial_z \frac{\sqrt{b\,f[z]^k}}{\left(\sqrt{f[z]^j}\right)^{j\,k}} = 0$
- **Rule h2: If** $j^2 = k^2 = 1 \bigwedge m - \frac{1}{2} \in \mathbb{Z} \bigwedge n - \frac{1}{2} \in \mathbb{Z} \bigwedge n > 0$**, then**

$$\int \left(\text{Sin}[c+d\,x]^j\right)^m \left(A+B\,\text{Sin}[c+d\,x]^k+C\,\text{Sin}[c+d\,x]^{2\,k}\right) \left(b\,\text{Sin}[c+d\,x]^k\right)^n dx \longrightarrow$$
$$\frac{b^{n-\frac{1}{2}} \sqrt{b\,\text{Sin}[c+d\,x]^k}}{\left(\sqrt{\text{Sin}[c+d\,x]^j}\right)^{j\,k}} \int \text{Sin}[c+d\,x]^{j\,m+k\,n} \left(A+B\,\text{Sin}[c+d\,x]^k+C\,\text{Sin}[c+d\,x]^{2\,k}\right) dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_*(A_.+B_.*sin[c_.+d_.*x_]^k_.+C_.*sin[c_.+d_.*x_]^k2_)*
    (b_*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  Dist[b^(n-1/2)*Sqrt[b*Sin[c+d*x]^k]/(Sqrt[Sin[c+d*x]^j])^(j*k),
    Int[sin[c+d*x]^(j*m+k*n)*(A+B*sin[c+d*x]^k+C*sin[c+d*x]^(2*k)),x]] /;
FreeQ[{b,c,d,A,B,C},x] && OneQ[j^2,k^2] && IntegerQ[m-1/2] && IntegerQ[n-1/2] && n>0
```

```
Int[(sin[c_.+d_.*x_]^j_.)^m_*(A_.+C_.*sin[c_.+d_.*x_]^k2_)*(b_*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  Dist[b^(n-1/2)*Sqrt[b*Sin[c+d*x]^k]/(Sqrt[Sin[c+d*x]^j])^(j*k),
    Int[sin[c+d*x]^(j*m+k*n)*(A+C*sin[c+d*x]^(2*k)),x]] /;
FreeQ[{b,c,d,A,C},x] && OneQ[j^2,k^2] && IntegerQ[m-1/2] && IntegerQ[n-1/2] && n>0
```

- **Derivation: Piecewise constant extraction**
- **Basis: If** $j^2 = 1$**, then** $\partial_z \frac{\left(\sqrt{f[z]^j}\right)^{j k}}{\sqrt{b f[z]^k}} = 0$
- **Rule h3: If** $j^2 = k^2 = 1 \bigwedge m - \frac{1}{2} \in \mathbb{Z} \bigwedge n - \frac{1}{2} \in \mathbb{Z} \bigwedge n < 0$**, then**

$$\int \left(\text{Sin}[c + d x]^j\right)^m \left(A + B \,\text{Sin}[c + d x]^k + C \,\text{Sin}[c + d x]^{2 k}\right) \left(b \,\text{Sin}[c + d x]^k\right)^n dx \longrightarrow$$

$$\frac{b^{n+\frac{1}{2}} \left(\sqrt{\text{Sin}[c + d x]^j}\right)^{j k}}{\sqrt{b \,\text{Sin}[c + d x]^k}} \int \text{Sin}[c + d x]^{j m+k n} \left(A + B \,\text{Sin}[c + d x]^k + C \,\text{Sin}[c + d x]^{2 k}\right) dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_*(A_.+B_.*sin[c_.+d_.*x_]^k_.+C_.*sin[c_.+d_.*x_]^k2_)*
    (b_*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  Dist[b^(n+1/2)*(Sqrt[Sin[c+d*x]^j])^(j*k)/Sqrt[b*Sin[c+d*x]^k],
    Int[sin[c+d*x]^(j*m+k*n)*(A+B*sin[c+d*x]^k+C*sin[c+d*x]^(2*k)),x]] /;
FreeQ[{b,c,d,A,B,C},x] && OneQ[j^2,k^2] && IntegerQ[m-1/2] && IntegerQ[n-1/2] && n<0
```

```
Int[(sin[c_.+d_.*x_]^j_.)^m_*(A_.+C_.*sin[c_.+d_.*x_]^k2_)*(b_*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  Dist[b^(n+1/2)*(Sqrt[Sin[c+d*x]^j])^(j*k)/Sqrt[b*Sin[c+d*x]^k],
    Int[sin[c+d*x]^(j*m+k*n)*(A+C*sin[c+d*x]^(2*k)),x]] /;
FreeQ[{b,c,d,A,C},x] && OneQ[j^2,k^2] && IntegerQ[m-1/2] && IntegerQ[n-1/2] && n<0
```

Rule i:

$$\int \left(\text{Sin}[c+d\,x]^j\right)^m \left(A+B\,\text{Csc}[c+d\,x]+C\,\text{Csc}[c+d\,x]^2\right)\,(a+b\,\text{Csc}[c+d\,x])^n\,dx$$

- **Derivation: Algebraic simplification**
- **Rule i1: If** $j^2=1\wedge a^2-b^2\neq 0\wedge -1<m\leq 1$, **then**

$$\int \frac{\left(\text{Sin}[c+d\,x]^j\right)^m \left(A+B\,\text{Csc}[c+d\,x]+C\,\text{Csc}[c+d\,x]^2\right)}{a+b\,\text{Csc}[c+d\,x]}\,dx \longrightarrow$$
$$\int \frac{\left(\text{Sin}[c+d\,x]^j\right)^{m-j} \left(C+B\,\text{Sin}[c+d\,x]+A\,\text{Sin}[c+d\,x]^2\right)}{b+a\,\text{Sin}[c+d\,x]}\,dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(A_.+B_.*sin[c_.+d_.*x_]^(-1)+C_.*sin[c_.+d_.*x_]^(-2))/
    (a_+b_.*sin[c_.+d_.*x_]^(-1)),x_Symbol] :=
  Int[(sin[c+d*x]^j)^(m-j)*(C+B*sin[c+d*x]+A*sin[c+d*x]^2)/(b+a*sin[c+d*x]),x] /;
FreeQ[{a,b,c,d,A,B,C},x] && OneQ[j^2] && NonzeroQ[a^2-b^2] && RationalQ[m] && -1<m≤1
```

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(A_.+C_.*sin[c_.+d_.*x_]^(-2))/(a_+b_.*sin[c_.+d_.*x_]^(-1)),x_Symbol]
  Int[(sin[c+d*x]^j)^(m-j)*(C+A*sin[c+d*x]^2)/(b+a*sin[c+d*x]),x] /;
FreeQ[{a,b,c,d,A,C},x] && OneQ[j^2] && NonzeroQ[a^2-b^2] && RationalQ[m] && -1<m≤1
```

- **Derivation: Piecewise constant extraction**
- **Basis:** $\partial_z \frac{\sqrt{b+a\,f[z]}}{\sqrt{f[z]}\,\sqrt{a+b/f[z]}}=0$
- **Rule i2: If** $a^2-b^2\neq 0$, **then**

$$\int \frac{\text{Sin}[c+d\,x]\left(A+B\,\text{Csc}[c+d\,x]+C\,\text{Csc}[c+d\,x]^2\right)}{\sqrt{a+b\,\text{Csc}[c+d\,x]}}\,dx \longrightarrow$$
$$\frac{\sqrt{b+a\,\text{Sin}[c+d\,x]}}{\sqrt{\text{Sin}[c+d\,x]}\,\sqrt{a+b\,\text{Csc}[c+d\,x]}}\int \frac{C+B\,\text{Sin}[c+d\,x]+A\,\text{Sin}[c+d\,x]^2}{\sqrt{\text{Sin}[c+d\,x]}\,\sqrt{b+a\,\text{Sin}[c+d\,x]}}\,dx$$

- **Program code:**

```
Int[sin[c_.+d_.*x_]*(A_.+B_.*sin[c_.+d_.*x_]^(-1)+C_.*sin[c_.+d_.*x_]^(-2))/
    Sqrt[a_.+b_.*sin[c_.+d_.*x_]^(-1)],x_Symbol] :=
  Dist[Sqrt[b+a*Sin[c+d*x]]/(Sqrt[Sin[c+d*x]]*Sqrt[a+b*Csc[c+d*x]]),
    Int[(C+B*sin[c+d*x]+A*sin[c+d*x]^2)/(Sqrt[sin[c+d*x]]*Sqrt[b+a*sin[c+d*x]]),x]] /;
FreeQ[{a,b,c,d,A,B,C},x] && NonzeroQ[a^2-b^2]
```

```
Int[sin[c_.+d_.*x_]*(A_.+C_.*sin[c_.+d_.*x_]^(-2))/Sqrt[a_.+b_.*sin[c_.+d_.*x_]^(-1)],x_Symbol] :=
  Dist[Sqrt[b+a*Sin[c+d*x]]/(Sqrt[Sin[c+d*x]]*Sqrt[a+b*Csc[c+d*x]]),
    Int[(C+A*sin[c+d*x]^2)/(Sqrt[sin[c+d*x]]*Sqrt[b+a*sin[c+d*x]]),x]] /;
FreeQ[{a,b,c,d,A,C},x] && NonzeroQ[a^2-b^2]
```

- **Derivation: Piecewise constant extraction**
- **Basis: If** $j^2 = 1$, **then** $\partial_z \frac{\sqrt{b+a f[z]}}{\left(\sqrt{f[z]^j}\right)^j \sqrt{a+b f[z]^{-1}}} = 0$
- **Rule i3: If** $j^2 = 1 \wedge a^2 - b^2 \neq 0 \wedge j m = \frac{1}{2}$, **then**

$$\int \frac{\left(\text{Sin}[c+d x]^j\right)^m \left(A + B\,\text{Csc}[c+d x] + C\,\text{Csc}[c+d x]^2\right)}{\sqrt{a + b\,\text{Csc}[c+d x]}}\, dx \longrightarrow$$

$$\frac{\sqrt{b + a\,\text{Sin}[c+d x]}}{\sqrt{\text{Sin}[c+d x]^j}^{\,j} \sqrt{a + b\,\text{Csc}[c+d x]}} \int \frac{\text{Sin}[c+d x]^{j m-3/2} \left(C + B\,\text{Sin}[c+d x] + A\,\text{Sin}[c+d x]^2\right)}{\sqrt{b + a\,\text{Sin}[c+d x]}}\, dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_*(A_.+B_.*sin[c_.+d_.*x_]^(-1)+C_.*sin[c_.+d_.*x_]^(-2))/
    Sqrt[a_.+b_.*sin[c_.+d_.*x_]^(-1)],x_Symbol] :=
  Dist[Sqrt[b+a*Sin[c+d*x]]/((Sqrt[Sin[c+d*x]^j])^j*Sqrt[a+b*Csc[c+d*x]]),
    Int[sin[c+d*x]^(j*m-3/2)*(C+B*sin[c+d*x]+A*sin[c+d*x]^2)/Sqrt[b+a*sin[c+d*x]],x]] /;
FreeQ[{a,b,c,d,A,B,C},x] && OneQ[j^2] && NonzeroQ[a^2-b^2] && ZeroQ[j*m-1/2]
```

```
Int[(sin[c_.+d_.*x_]^j_.)^m_*(A_.+C_.*sin[c_.+d_.*x_]^(-2))/Sqrt[a_.+b_.*sin[c_.+d_.*x_]^(-1)],x_Sym
  Dist[Sqrt[b+a*Sin[c+d*x]]/((Sqrt[Sin[c+d*x]^j])^j*Sqrt[a+b*Csc[c+d*x]]),
    Int[sin[c+d*x]^(j*m-3/2)*(C+A*sin[c+d*x]^2)/Sqrt[b+a*sin[c+d*x]],x]] /;
FreeQ[{a,b,c,d,A,C},x] && OneQ[j^2] && NonzeroQ[a^2-b^2] && ZeroQ[j*m-1/2]
```

Rule

$$\text{j:}\quad \int \mathrm{Csc}[c+d\,x]^m \left(A+B\,\mathrm{Sin}[c+d\,x]+C\,\mathrm{Sin}[c+d\,x]^2\right)(a+b\,\mathrm{Sin}[c+d\,x])^n\,dx$$

- **Derivation: Piecewise constant extraction**
- **Basis:** $\partial_z\left(\mathrm{Sin}[z]^m\,\mathrm{Csc}[z]^m\right)=0$
- **Rule j: If** $m-\frac{1}{2}\in\mathbb{Z}\ \bigwedge\ 0<m<2\ \bigwedge\ -2<n<0$**, then**

$$\int \mathrm{Csc}[c+d\,x]^m \left(A+B\,\mathrm{Sin}[c+d\,x]+C\,\mathrm{Sin}[c+d\,x]^2\right)(a+b\,\mathrm{Sin}[c+d\,x])^n\,dx\ \longrightarrow$$

$$\sqrt{\mathrm{Csc}[c+d\,x]}\ \sqrt{\mathrm{Sin}[c+d\,x]}\int\frac{\left(A+B\,\mathrm{Sin}[c+d\,x]+C\,\mathrm{Sin}[c+d\,x]^2\right)(a+b\,\mathrm{Sin}[c+d\,x])^n}{\mathrm{Sin}[c+d\,x]^m}\,dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^(-1))^m_*(A_.+B_.*sin[c_.+d_.*x_]+C_.*sin[c_.+d_.*x_]^2)*
    (a_+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  Dist[Sqrt[Csc[c+d*x]]*Sqrt[Sin[c+d*x]],
    Int[(A+B*sin[c+d*x]+C*sin[c+d*x]^2)*(a+b*sin[c+d*x])^n/sin[c+d*x]^m,x]] /;
FreeQ[{a,b,c,d,A,B,C},x] && IntegerQ[m-1/2] && RationalQ[n] && 0<m<2 && -2<n<0
```

```
Int[(sin[c_.+d_.*x_]^(-1))^m_*(A_.+C_.*sin[c_.+d_.*x_]^2)*(a_+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  Dist[Sqrt[Csc[c+d*x]]*Sqrt[Sin[c+d*x]],
    Int[(A+C*sin[c+d*x]^2)*(a+b*sin[c+d*x])^n/sin[c+d*x]^m,x]] /;
FreeQ[{a,b,c,d,A,C},x] && IntegerQ[m-1/2] && RationalQ[n] && 0<m<2 && -2<n<0
```

Rules 9 – 10:

$$\int \mathrm{Sin}[c+d\,x]^{\frac{k-1}{2}}\left(A+B\,\mathrm{Sin}[c+d\,x]^k+C\,\mathrm{Sin}[c+d\,x]^{2k}\right)\left(a+b\,\mathrm{Sin}[c+d\,x]^k\right)^n\,dx$$

- **Derivation: Rule 1 with** $m=\frac{k-1}{2}$
- **Rule 9: If** $k^2=1\wedge a^2-b^2\neq 0\wedge a^2 C-a\,b\,B+b^2 A\neq 0\wedge n<-1$**, then**

$$\int \mathrm{Sin}[c+d\,x]^{\frac{k-1}{2}}\left(A+B\,\mathrm{Sin}[c+d\,x]^k+C\,\mathrm{Sin}[c+d\,x]^{2k}\right)\left(a+b\,\mathrm{Sin}[c+d\,x]^k\right)^n\,dx\longrightarrow$$

$$-\frac{\left(a^2 C-a\,b\,B+b^2 A\right)\mathrm{Cos}[c+d\,x]\,\mathrm{Sin}[c+d\,x]^{\frac{k-1}{2}}\left(a+b\,\mathrm{Sin}[c+d\,x]^k\right)^{n+1}}{b\,d\,(n+1)\left(a^2-b^2\right)}+\frac{1}{b\,(n+1)\left(a^2-b^2\right)}\cdot$$

$$\int \mathrm{Sin}[c+d\,x]^{\frac{k-1}{2}}\left(b\,(a\,A-b\,B+a\,C)\,(n+1)-\left(a^2 C-a\,b\,B+b^2 A+\left(b^2 A-a\,b\,B+b^2 C\right)(n+1)\right)\mathrm{Sin}[c+d\,x]^k\right)$$
$$\left(a+b\,\mathrm{Sin}[c+d\,x]^k\right)^{n+1}\,dx$$

- **Program code:**

```
Int[(A_.+B_.*sin[c_.+d_.*x_]+C_.*sin[c_.+d_.*x_]^2)*(a_+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  -(a^2*C-a*b*B+b^2*A)*Cos[c+d*x]*(a+b*Sin[c+d*x])^(n+1)/(b*d*(n+1)*(a^2-b^2)) +
  Dist[1/(b*(n+1)*(a^2-b^2)),
    Int[Sim[b*(a*A-b*B+a*C)*(n+1)-(a^2*C-a*b*B+b^2*A+(b^2*A-a*b*B+b^2*C)*(n+1))*sin[c+d*x],x]*
      (a+b*sin[c+d*x])^(n+1),x]] /;
FreeQ[{a,b,c,d,A,B,C},x] && NonzeroQ[a^2-b^2] && NonzeroQ[a^2*C-a*b*B+b^2*A] && RationalQ[n] && n<-1
```

```
Int[(A_.+C_.*sin[c_.+d_.*x_]^2)*(a_+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  -(a^2*C+b^2*A)*Cos[c+d*x]*(a+b*Sin[c+d*x])^(n+1)/(b*d*(n+1)*(a^2-b^2)) +
  Dist[1/(b*(n+1)*(a^2-b^2)),
    Int[Sim[a*b*(A+C)*(n+1)-(a^2*C+b^2*A+(b^2*A+b^2*C)*(n+1))*sin[c+d*x],x]*
      (a+b*sin[c+d*x])^(n+1),x]] /;
FreeQ[{a,b,c,d,A,C},x] && NonzeroQ[a^2-b^2] && NonzeroQ[a^2*C+b^2*A] && RationalQ[n] && n<-1
```

```
Int[sin[c_.+d_.*x_]^(-1)*(A_.+B_.*sin[c_.+d_.*x_]^(-1)+C_.*sin[c_.+d_.*x_]^(-2))*
    (a_+b_.*sin[c_.+d_.*x_]^(-1))^n_,x_Symbol] :=
  -(a^2*C-a*b*B+b^2*A)*Cot[c+d*x]*(a+b*Csc[c+d*x])^(n+1)/(b*d*(n+1)*(a^2-b^2)) +
  Dist[1/(b*(n+1)*(a^2-b^2)),
    Int[sin[c+d*x]^(-1)*
      Sim[b*(a*A-b*B+a*C)*(n+1)-(a^2*C-a*b*B+b^2*A+(b^2*A-a*b*B+b^2*C)*(n+1))*sin[c+d*x]^(-1),x]*
      (a+b*sin[c+d*x]^(-1))^(n+1),x]] /;
FreeQ[{a,b,c,d,A,B,C},x] && NonzeroQ[a^2-b^2] && NonzeroQ[a^2*C-a*b*B+b^2*A] && RationalQ[n] && n<-1
```

```
Int[sin[c_.+d_.*x_]^(-1)*(A_.+C_.*sin[c_.+d_.*x_]^(-2))*(a_+b_.*sin[c_.+d_.*x_]^(-1))^n_,x_Symbol] :
  -(a^2*C+b^2*A)*Cot[c+d*x]*(a+b*Csc[c+d*x])^(n+1)/(b*d*(n+1)*(a^2-b^2)) +
  Dist[1/(b*(n+1)*(a^2-b^2)),
    Int[sin[c+d*x]^(-1)*
      Sim[a*b*(A+C)*(n+1)-(a^2*C+b^2*A+(b^2*A+b^2*C)*(n+1))*sin[c+d*x]^(-1),x]*
      (a+b*sin[c+d*x]^(-1))^(n+1),x]] /;
FreeQ[{a,b,c,d,A,C},x] && NonzeroQ[a^2-b^2] && NonzeroQ[a^2*C+b^2*A] && RationalQ[n] && n<-1
```

- **Derivation: Rule 2 with** $m = \frac{k-1}{2}$

- **Rule 10: If** $k^2 = 1 \wedge n > -1$**, then**

$$\int \text{Sin}[c + d\,x]^{\frac{k-1}{2}} \left(A + B\,\text{Sin}[c + d\,x]^k + C\,\text{Sin}[c + d\,x]^{2\,k}\right) \left(a + b\,\text{Sin}[c + d\,x]^k\right)^n dx \longrightarrow$$

$$-\frac{C\,\text{Cos}[c + d\,x]\,\text{Sin}[c + d\,x]^{\frac{k-1}{2}} \left(a + b\,\text{Sin}[c + d\,x]^k\right)^{n+1}}{b\,d\,(n + 2)} + \frac{1}{b\,(n + 2)} \cdot$$

$$\int \text{Sin}[c + d\,x]^{\frac{k-1}{2}} \left(b\,A\,(n + 2) + b\,C\,(n + 1) + (b\,B\,(n + 2) - a\,C)\,\text{Sin}[c + d\,x]^k\right) \left(a + b\,\text{Sin}[c + d\,x]^k\right)^n dx$$

- **Program code:**

```
Int[(A_.+B_.*sin[c_.+d_.*x_]+C_.*sin[c_.+d_.*x_]^2)*(a_.+b_.*sin[c_.+d_.*x_])^n_.,x_Symbol] :=
  -C*Cos[c+d*x]*(a+b*Sin[c+d*x])^(n+1)/(b*d*(n+2)) +
  Dist[1/(b*(n+2)),
    Int[Sim[b*A*(n+2)+b*C*(n+1)+(b*B*(n+2)-a*C)*sin[c+d*x],x]*(a+b*sin[c+d*x])^n,x]] /;
FreeQ[{a,b,c,d,A,B,C},x] && RationalQ[n] && n>-1
```

```
Int[(A_.+C_.*sin[c_.+d_.*x_]^2)*(a_.+b_.*sin[c_.+d_.*x_])^n_.,x_Symbol] :=
  -C*Cos[c+d*x]*(a+b*Sin[c+d*x])^(n+1)/(b*d*(n+2)) +
  Dist[1/(b*(n+2)),
    Int[Sim[b*A*(n+2)+b*C*(n+1)-a*C*sin[c+d*x],x]*(a+b*sin[c+d*x])^n,x]] /;
FreeQ[{a,b,c,d,A,C},x] && RationalQ[n] && n>-1
```

```
Int[sin[c_.+d_.*x_]^(-1)*(A_.+B_.*sin[c_.+d_.*x_]^(-1)+C_.*sin[c_.+d_.*x_]^(-2))*
    (a_+b_.*sin[c_.+d_.*x_]^(-1))^n_.,x_Symbol] :=
  -C*Cot[c+d*x]*(a+b*Csc[c+d*x])^(n+1)/(b*d*(n+2)) +
  Dist[1/(b*(n+2)),
    Int[sin[c+d*x]^(-1)*
      Sim[b*A*(n+2)+b*C*(n+1)+(b*B*(n+2)-a*C)*sin[c+d*x]^(-1),x]*(a+b*sin[c+d*x]^(-1))^n,x]] /;
FreeQ[{a,b,c,d,A,B,C},x] && RationalQ[n] && n>-1
```

```
Int[sin[c_.+d_.*x_]^(-1)*(A_.+C_.*sin[c_.+d_.*x_]^(-2))*(a_+b_.*sin[c_.+d_.*x_]^(-1))^n_.,x_Symbol]
  -C*Cot[c+d*x]*(a+b*Csc[c+d*x])^(n+1)/(b*d*(n+2)) +
  Dist[1/(b*(n+2)),
    Int[sin[c+d*x]^(-1)*
      Sim[b*A*(n+2)+b*C*(n+1)-a*C*sin[c+d*x]^(-1),x]*(a+b*sin[c+d*x]^(-1))^n,x]] /;
FreeQ[{a,b,c,d,A,C},x] && RationalQ[n] && n>-1
```

Rules 11 – 12:

$$\int \left(\text{Sin}[c+d\,x]^j\right)^m \left(A+B\,\text{Sin}[c+d\,x]^k+C\,\text{Sin}[c+d\,x]^{2\,k}\right) \left(a+b\,\text{Sin}[c+d\,x]^k\right) dx$$

- **Note: The rules in this section would only generate slightly simpler antiderivatives and require as many steps as using rules 3 and 4 directly.**
- **Derivation: Rule 4 with $n = 1$ and ???**
- **Rule 11: If $j^2 = k^2 = 1 \wedge j\,k\,m < -1$, then**

$$\int \left(\text{Sin}[c+d\,x]^j\right)^m \left(A+B\,\text{Sin}[c+d\,x]^k+C\,\text{Sin}[c+d\,x]^{2\,k}\right) \left(a+b\,\text{Sin}[c+d\,x]^k\right) dx \longrightarrow$$

$$\frac{a\,A\,\text{Cos}[c+d\,x]\,\left(\text{Sin}[c+d\,x]^j\right)^{m+j\,k}}{d\left(j\,k\,m+\frac{k+1}{2}\right)}+\frac{1}{j\,k\,m+\frac{k+1}{2}}\int \left(\text{Sin}[c+d\,x]^j\right)^{m+j\,k}\cdot$$
$$\left(\left(j\,k\,m+\frac{k+1}{2}\right)(b\,A+a\,B)+\right.$$
$$\left.\left(\left(j\,k\,m+\frac{k+3}{2}\right)a\,A+\left(j\,k\,m+\frac{k+1}{2}\right)(b\,B+a\,C)\right)\text{Sin}[c+d\,x]^k+\left(j\,k\,m+\frac{k+1}{2}\right)b\,C\,\text{Sin}[c+d\,x]^{2\,k}\right)dx$$

- **Program code:**

```
(* Int[(sin[c_.+d_.*x_]^j_.)^m_.*(A_.+B_.*sin[c_.+d_.*x_]^k_.+C_.*sin[c_.+d_.*x_]^k2_)*
    (a_+b_.*sin[c_.+d_.*x_]^k_.),x_Symbol] :=
  a*A*Cos[c+d*x]*(Sin[c+d*x]^j)^(m+j*k)/(d*(j*k*m+(k+1)/2)) +
  Dist[1/(j*k*m+(k+1)/2),
    Int[(sin[c+d*x]^j)^(m+j*k)*
      Sim[(j*k*m+(k+1)/2)*(b*A+a*B)+((j*k*m+(k+3)/2)*a*A+(j*k*m+(k+1)/2)*(b*B+a*C))*sin[c+d*x]^k+
        (j*k*m+(k+1)/2)*b*C*sin[c+d*x]^(2*k),x],x]] /;
FreeQ[{a,b,c,d,A,B,C},x] && OneQ[j^2,k^2] && k2===2*k && RationalQ[m] && j*k*m<-1 *)
```

```
(* Int[(sin[c_.+d_.*x_]^j_.)^m_.*(A_.+C_.*sin[c_.+d_.*x_]^k2_)*
    (a_+b_.*sin[c_.+d_.*x_]^k_.),x_Symbol] :=
  a*A*Cos[c+d*x]*(Sin[c+d*x]^j)^(m+j*k)/(d*(j*k*m+(k+1)/2)) +
  Dist[1/(j*k*m+(k+1)/2),
    Int[(sin[c+d*x]^j)^(m+j*k)*
      Sim[(j*k*m+(k+1)/2)*b*A+((j*k*m+(k+3)/2)*a*A+(j*k*m+(k+1)/2)*a*C)*sin[c+d*x]^k+
        (j*k*m+(k+1)/2)*b*C*sin[c+d*x]^(2*k),x],x]] /;
FreeQ[{a,b,c,d,A,C},x] && OneQ[j^2,k^2] && k2===2*k && RationalQ[m] && j*k*m<-1 *)
```

- **Derivation: Rule 3 with $n = 1$ and ???**

- **Rule 12: If $j^2 = k^2 = 1 \wedge j\,k\,m \geq -1$, then**

$$\int \left(\text{Sin}[c+d\,x]^j\right)^m \left(A+B\,\text{Sin}[c+d\,x]^k+C\,\text{Sin}[c+d\,x]^{2\,k}\right) \left(a+b\,\text{Sin}[c+d\,x]^k\right) dx \longrightarrow$$

$$-\frac{b\,C\,\text{Cos}[c+d\,x] \left(\text{Sin}[c+d\,x]^j\right)^{m+2\,j\,k}}{d\left(j\,k\,m+\frac{k+5}{2}\right)} + \frac{1}{j\,k\,m+\frac{k+5}{2}} \int \left(\text{Sin}[c+d\,x]^j\right)^m \cdot$$
$$\left(\left(j\,k\,m+\frac{k+5}{2}\right) a\,A + \left(\left(j\,k\,m+\frac{k+5}{2}\right)(b\,A+a\,B) + \left(j\,k\,m+\frac{k+3}{2}\right) b\,C\right) \text{Sin}[c+d\,x]^k + \left(j\,k\,m+\frac{k+5}{2}\right)(b\,B+a\,C)\,\text{Sin}[c+d\,x]^{2\,k}\right) dx$$

- **Program code:**

```
(* Int[(sin[c_.+d_.*x_]^j_.)^m_.*(A_.+B_.*sin[c_.+d_.*x_]^k_.+C_.*sin[c_.+d_.*x_]^k2_)*
    (a_+b_.*sin[c_.+d_.*x_]^k_.),x_Symbol] :=
  -b*C*Cos[c+d*x]*(Sin[c+d*x]^j)^(m+2*j*k)/(d*(j*k*m+(k+5)/2)) +
  Dist[1/(j*k*m+(k+5)/2),
    Int[(sin[c+d*x]^j)^m*
      Sim[(j*k*m+(k+5)/2)*a*A+((j*k*m+(k+5)/2)*(b*A+a*B)+(j*k*m+(k+3)/2)*b*C)*sin[c+d*x]^k+
        (j*k*m+(k+5)/2)*(b*B+a*C)*sin[c+d*x]^(2*k),x],x]] /;
FreeQ[{a,b,c,d,A,B,C},x] && OneQ[j^2,k^2] && k2===2*k && RationalQ[m] && j*k*m>=-1 *)
```

```
(* Int[(sin[c_.+d_.*x_]^j_.)^m_.*(A_.+C_.*sin[c_.+d_.*x_]^k2_)*
    (a_+b_.*sin[c_.+d_.*x_]^k_.),x_Symbol] :=
  -b*C*Cos[c+d*x]*(Sin[c+d*x]^j)^(m+2*j*k)/(d*(j*k*m+(k+5)/2)) +
  Dist[1/(j*k*m+(k+5)/2),
    Int[(sin[c+d*x]^j)^m*
      Sim[(j*k*m+(k+5)/2)*a*A+((j*k*m+(k+5)/2)*b*A+(j*k*m+(k+3)/2)*b*C)*sin[c+d*x]^k+
        (j*k*m+(k+5)/2)*a*C*sin[c+d*x]^(2*k),x],x]] /;
FreeQ[{a,b,c,d,A,C},x] && OneQ[j^2,k^2] && k2===2*k && RationalQ[m] && j*k*m>=-1 *)
```

Rules 1 – 6: $\int \left(\text{Sin}[c+d\,x]^j\right)^m \left(A+B\,\text{Sin}[c+d\,x]^k+C\,\text{Sin}[c+d\,x]^{2\,k}\right)\left(a+b\,\text{Sin}[c+d\,x]^k\right)^n \, dx$

- **Derivation: Rule 1 or 6 with** $a^2\,C-a\,b\,B+b^2\,A=0$
- **Derivation: Algebraic simplification**
- **Basis: If** $a^2\,C-a\,b\,B+b^2\,A=0$, **then** $A+B\,z+C\,z^2=\frac{(b\,B-a\,C+b\,C\,z)\,(a+b\,z)}{b^2}$
- **Rule: If** $j^2=k^2=1\,\land\,a^2\,C-a\,b\,B+b^2\,A=0\,\land\,n<-1$, **then**

$$\int \left(\text{Sin}[c+d\,x]^j\right)^m \left(A+B\,\text{Sin}[c+d\,x]^k+C\,\text{Sin}[c+d\,x]^{2\,k}\right)\left(a+b\,\text{Sin}[c+d\,x]^k\right)^n dx \longrightarrow$$
$$\frac{1}{b^2}\int \left(\text{Sin}[c+d\,x]^j\right)^m \left(b\,B-a\,C+b\,C\,\text{Sin}[c+d\,x]^k\right)\left(a+b\,\text{Sin}[c+d\,x]^k\right)^{n+1} dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(A_.+B_.*sin[c_.+d_.*x_]^k_.+C_.*sin[c_.+d_.*x_]^k2_)*
    (a_+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  Dist[1/b^2,Int[(sin[c+d*x]^j)^m*Sim[b*B-a*C+b*C*sin[c+d*x]^k,x]*(a+b*sin[c+d*x]^k)^(n+1),x]] /;
FreeQ[{a,b,c,d,A,B,C,m},x] && OneQ[j^2,k^2] && k2===2*k && ZeroQ[a^2*C-a*b*B+b^2*A] && RationalQ[n]
```

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(A_.+C_.*sin[c_.+d_.*x_]^k2_)*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol
  Dist[1/b^2,Int[(sin[c+d*x]^j)^m*Sim[-a*C+b*C*sin[c+d*x]^k,x]*(a+b*sin[c+d*x]^k)^(n+1),x]] /;
FreeQ[{a,b,c,d,A,C,m},x] && OneQ[j^2,k^2] && k2===2*k && ZeroQ[a^2*C+b^2*A] && RationalQ[n] && n<-1
```

- **Derivation: Recurrence 1**

- **Rule 1: If** $j^2 = k^2 = 1 \bigwedge a^2 - b^2 \neq 0 \bigwedge a^2 C - a b B + b^2 A \neq 0 \bigwedge j k m > 0 \bigwedge n < -1$**, then**

$$\int \left(\text{Sin}[c+d x]^j\right)^m \left(A + B\,\text{Sin}[c+d x]^k + C\,\text{Sin}[c+d x]^{2k}\right) \left(a + b\,\text{Sin}[c+d x]^k\right)^n \, dx \longrightarrow$$

$$-\frac{\left(a^2 C - a b B + b^2 A\right) \text{Cos}[c+d x] \left(\text{Sin}[c+d x]^j\right)^m \left(a + b\,\text{Sin}[c+d x]^k\right)^{n+1}}{b d (n+1) \left(a^2 - b^2\right)} +$$

$$\frac{1}{b (n+1) \left(a^2 - b^2\right)} \int \left(\text{Sin}[c+d x]^j\right)^{m-j k} \cdot$$

$$\left(\left(a^2 C - a b B + b^2 A\right) \left(j k m + \frac{k-1}{2}\right) + b (a A - b B + a C) (n+1) \text{Sin}[c+d x]^k - \right.$$

$$\left(\left(b^2 A - a b B + b^2 C\right) (n+1) + \left(a^2 C - a b B + b^2 A\right) \left(j k m + \frac{k+1}{2}\right)\right)$$

$$\left.\text{Sin}[c+d x]^{2k}\right) \left(a + b\,\text{Sin}[c+d x]^k\right)^{n+1} dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(A_.+B_.*sin[c_.+d_.*x_]^k_.+C_.*sin[c_.+d_.*x_]^k2_)*
    (a_+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  -(a^2*C-a*b*B+b^2*A)*Cos[c+d*x]*(Sin[c+d*x]^j)^m*(a+b*Sin[c+d*x]^k)^(n+1)/(b*d*(n+1)*(a^2-b^2)) +
  Dist[1/(b*(n+1)*(a^2-b^2)),
    Int[(sin[c+d*x]^j)^(m-j*k)*
      Sim[(a^2*C-a*b*B+b^2*A)*(j*k*m+(k-1)/2)+b*(a*A-b*B+a*C)*(n+1)*sin[c+d*x]^k-
        ((b^2*A-a*b*B+b^2*C)*(n+1)+(a^2*C-a*b*B+b^2*A)*(j*k*m+(k+1)/2))*sin[c+d*x]^(2*k),x]*
      (a+b*sin[c+d*x]^k)^(n+1),x]] /;
FreeQ[{a,b,c,d,A,B,C},x] && OneQ[j^2,k^2] && k2===2*k && NonzeroQ[a^2-b^2] &&
  NonzeroQ[a^2*C-a*b*B+b^2*A] && RationalQ[m,n] && j*k*m>0 && n<-1
```

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(A_.+C_.*sin[c_.+d_.*x_]^k2_)*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol
  -(a^2*C+b^2*A)*Cos[c+d*x]*(Sin[c+d*x]^j)^m*(a+b*Sin[c+d*x]^k)^(n+1)/(b*d*(n+1)*(a^2-b^2)) +
  Dist[1/(b*(n+1)*(a^2-b^2)),
    Int[(sin[c+d*x]^j)^(m-j*k)*
      Sim[(a^2*C+b^2*A)*(j*k*m+(k-1)/2)+b*(a*A+a*C)*(n+1)*sin[c+d*x]^k-
        ((b^2*A+b^2*C)*(n+1)+(a^2*C+b^2*A)*(j*k*m+(k+1)/2))*sin[c+d*x]^(2*k),x]*
      (a+b*sin[c+d*x]^k)^(n+1),x]] /;
FreeQ[{a,b,c,d,A,C},x] && OneQ[j^2,k^2] && k2===2*k && NonzeroQ[a^2-b^2] &&
  NonzeroQ[a^2*C+b^2*A] && RationalQ[m,n] && j*k*m>0 && n<-1
```

- **Derivation: Recurrence 2**
- **Rule 2: If** $j^2 = k^2 = 1 \wedge a^2 - b^2 \neq 0 \wedge j\,k\,m > 0 \wedge -1 \leq n < 0$, **then**

$$\int \left(\text{Sin}[c + d\,x]^j\right)^m \left(A + B\,\text{Sin}[c + d\,x]^k + C\,\text{Sin}[c + d\,x]^{2\,k}\right) \left(a + b\,\text{Sin}[c + d\,x]^k\right)^n \,dx \longrightarrow$$

$$-\frac{C\,\text{Cos}[c + d\,x] \left(\text{Sin}[c + d\,x]^j\right)^m \left(a + b\,\text{Sin}[c + d\,x]^k\right)^{n+1}}{b\,d\left(j\,k\,m + n + \frac{k+3}{2}\right)} + \frac{1}{b\left(j\,k\,m + n + \frac{k+3}{2}\right)} \int \left(\text{Sin}[c + d\,x]^j\right)^{m-j\,k} \cdot$$
$$\left(a\,C\left(j\,k\,m + \frac{k-1}{2}\right) + b\left(A + (A + C)\left(j\,k\,m + n + \frac{k+1}{2}\right)\right) \text{Sin}[c + d\,x]^k +\right.$$
$$\left.\left(b\,B\,(n + 1) + (b\,B - a\,C)\left(j\,k\,m + \frac{k+1}{2}\right)\right) \text{Sin}[c + d\,x]^{2\,k}\right) \left(a + b\,\text{Sin}[c + d\,x]^k\right)^n \,dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(A_.+B_.*sin[c_.+d_.*x_]^k_.+C_.*sin[c_.+d_.*x_]^k2_)*
    (a_+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  -C*Cos[c+d*x]*(Sin[c+d*x]^j)^m*(a+b*Sin[c+d*x]^k)^(n+1)/(b*d*(j*k*m+n+(k+3)/2)) +
  Dist[1/(b*(j*k*m+n+(k+3)/2)),
    Int[(sin[c+d*x]^j)^(m-j*k)*
      Sim[a*C*(j*k*m+(k-1)/2)+b*(A+(A+C)*(j*k*m+n+(k+1)/2))*sin[c+d*x]^k+
        (b*B*(n+1)+(b*B-a*C)*(j*k*m+(k+1)/2))*sin[c+d*x]^(2*k),x]*
      (a+b*sin[c+d*x]^k)^n,x]] /;
FreeQ[{a,b,c,d,A,B,C},x] && OneQ[j^2,k^2] && k2===2*k && NonzeroQ[a^2-b^2] &&
  RationalQ[m,n] && j*k*m>0 && -1≤n<0
```

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(A_.+C_.*sin[c_.+d_.*x_]^k2_)*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol
  -C*Cos[c+d*x]*(Sin[c+d*x]^j)^m*(a+b*Sin[c+d*x]^k)^(n+1)/(b*d*(j*k*m+n+(k+1)/2+1)) +
  Dist[1/(b*(j*k*m+n+(k+1)/2+1)),
    Int[(sin[c+d*x]^j)^(m-j*k)*
      Sim[a*C*(j*k*m+(k-1)/2)+b*(A+(A+C)*(j*k*m+n+(k+1)/2))*sin[c+d*x]^k-
        a*C*(j*k*m+(k+1)/2)*sin[c+d*x]^(2*k),x]*
      (a+b*sin[c+d*x]^k)^n,x]] /;
FreeQ[{a,b,c,d,A,C},x] && OneQ[j^2,k^2] && k2===2*k && NonzeroQ[a^2-b^2] &&
  RationalQ[m,n] && j*k*m>0 && -1≤n<0
```

- **Derivation: Recurrence 3**
- **Note: Rule 4 is used if $j\,k\,m = k = -1$.**
- **Rule 3: If $j^2 = k^2 = 1 \bigwedge a^2 - b^2 \neq 0 \bigwedge j\,k\,m \geq -1 \bigwedge \neg\,(m^2 = 1 \bigwedge k = -1) \bigwedge n > 0$, then**

$$\int \left(\text{Sin}[c+d\,x]^j\right)^m \left(A+B\,\text{Sin}[c+d\,x]^k+C\,\text{Sin}[c+d\,x]^{2\,k}\right) \left(a+b\,\text{Sin}[c+d\,x]^k\right)^n dx \longrightarrow$$

$$-\frac{C\,\text{Cos}[c+d\,x]\,\left(\text{Sin}[c+d\,x]^j\right)^{m+j\,k}\,\left(a+b\,\text{Sin}[c+d\,x]^k\right)^n}{d\,\left(j\,k\,m+\frac{k+3}{2}+n\right)}+\frac{1}{j\,k\,m+\frac{k+3}{2}+n}\int\left(\text{Sin}[c+d\,x]^j\right)^m \cdot$$
$$\left(a\left(A\,(n+1)+(A+C)\left(j\,k\,m+\frac{k+1}{2}\right)\right)+\left(b\,A+a\,B+(b\,A+a\,B+b\,C)\left(j\,k\,m+\frac{k+1}{2}+n\right)\right)\text{Sin}[c+d\,x]^k+\right.$$
$$\left.\left(a\,C\,n+b\,B\left(j\,k\,m+\frac{k+3}{2}+n\right)\right)\text{Sin}[c+d\,x]^{2\,k}\right)\left(a+b\,\text{Sin}[c+d\,x]^k\right)^{n-1}dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(A_.+B_.*sin[c_.+d_.*x_]^k_.+C_.*sin[c_.+d_.*x_]^k2_)*
    (a_+b_.*sin[c_.+d_.*x_]^k_.)^n_.,x_Symbol] :=
  -C*Cos[c+d*x]*(Sin[c+d*x]^j)^(m+j*k)*(a+b*Sin[c+d*x]^k)^n/(d*(j*k*m+(k+3)/2+n)) +
  Dist[1/(j*k*m+(k+3)/2+n),
    Int[(sin[c+d*x]^j)^m*
      Sim[a*(A*(n+1)+(A+C)*(j*k*m+(k+1)/2))+(b*A+a*B+(b*A+a*B+b*C)*(j*k*m+(k+1)/2+n))*sin[c+d*x]^k+
        (a*C*n+b*B*(j*k*m+(k+3)/2+n))*sin[c+d*x]^(2*k),x]*
      (a+b*sin[c+d*x]^k)^(n-1),x]] /;
FreeQ[{a,b,c,d,A,B,C},x] && OneQ[j^2,k^2] && k2===2*k && NonzeroQ[a^2-b^2] &&
  RationalQ[m,n] && j*k*m≥-1 && Not[m^2==1 && k==-1] && n>0
```

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(A_.+C_.*sin[c_.+d_.*x_]^k2_)*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_.,x_Symbo
  -C*Cos[c+d*x]*(Sin[c+d*x]^j)^(m+j*k)*(a+b*Sin[c+d*x]^k)^n/(d*(j*k*m+(k+3)/2+n)) +
  Dist[1/(j*k*m+(k+3)/2+n),
    Int[(sin[c+d*x]^j)^m*
      Sim[a*(A*(n+1)+(A+C)*(j*k*m+(k+1)/2))+(b*A+(b*A+b*C)*(j*k*m+(k+1)/2+n))*sin[c+d*x]^k+
        a*C*n*sin[c+d*x]^(2*k),x]*
      (a+b*sin[c+d*x]^k)^(n-1),x]] /;
FreeQ[{a,b,c,d,A,C},x] && OneQ[j^2,k^2] && k2===2*k && NonzeroQ[a^2-b^2] &&
  RationalQ[m,n] && j*k*m≥-1 && Not[m^2==1 && k==-1] && n>0
```

- **Derivation: Recurrence 4**

- **Rule 4: If** $j^2 = k^2 = 1 \bigwedge a^2 - b^2 \neq 0 \bigwedge j\,k\,m + \frac{k+1}{2} \neq 0 \bigwedge j\,k\,m \leq -1 \bigwedge n > 0$**, then**

$$\int \left(\text{Sin}[c+d\,x]^j\right)^m \left(A + B\,\text{Sin}[c+d\,x]^k + C\,\text{Sin}[c+d\,x]^{2\,k}\right) \left(a + b\,\text{Sin}[c+d\,x]^k\right)^n \,dx \longrightarrow$$

$$\frac{A\,\text{Cos}[c+d\,x] \left(\text{Sin}[c+d\,x]^j\right)^{m+j\,k} \left(a+b\,\text{Sin}[c+d\,x]^k\right)^n}{d\left(j\,k\,m+\frac{k+1}{2}\right)} + \frac{1}{j\,k\,m+\frac{k+1}{2}} \int \left(\text{Sin}[c+d\,x]^j\right)^{m+j\,k} \cdot$$
$$\left(a\,B\left(j\,k\,m+\frac{k+1}{2}\right) - b\,A\,n + \left(a\,A + (a\,A + a\,C + b\,B)\left(j\,k\,m+\frac{k+1}{2}\right)\right) \text{Sin}[c+d\,x]^k + \right.$$
$$\left. b\left(A\,(n+1) + (A+C)\left(j\,k\,m+\frac{k+1}{2}\right)\right) \text{Sin}[c+d\,x]^{2\,k}\right) \left(a+b\,\text{Sin}[c+d\,x]^k\right)^{n-1} dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(A_.+B_.*sin[c_.+d_.*x_]^k_.+C_.*sin[c_.+d_.*x_]^k2_)*
    (a_+b_.*sin[c_.+d_.*x_]^k_.)^n_.,x_Symbol] :=
  A*Cos[c+d*x]*(Sin[c+d*x]^j)^(m+j*k)*(a+b*Sin[c+d*x]^k)^n/(d*(j*k*m+(k+1)/2)) +
  Dist[1/(j*k*m+(k+1)/2),
     Int[(sin[c+d*x]^j)^(m+j*k)*
        Sim[a*B*(j*k*m+(k+1)/2)-b*A*n+(a*A+(a*A+a*C+b*B)*(j*k*m+(k+1)/2))*sin[c+d*x]^k+
          b*(A*(n+1)+(A+C)*(j*k*m+(k+1)/2))*sin[c+d*x]^(2*k),x]*
        (a+b*sin[c+d*x]^k)^(n-1),x]] /;
FreeQ[{a,b,c,d,A,B,C},x] && OneQ[j^2,k^2] && k2===2*k && NonzeroQ[a^2-b^2] &&
  RationalQ[m,n] && j*k*m+(k+1)/2≠0 && j*k*m≤-1 && n>0
```

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(A_.+C_.*sin[c_.+d_.*x_]^k2_)*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_.,x_Symbo
  A*Cos[c+d*x]*(Sin[c+d*x]^j)^(m+j*k)*(a+b*Sin[c+d*x]^k)^n/(d*(j*k*m+(k+1)/2)) +
  Dist[1/(j*k*m+(k+1)/2),
     Int[(sin[c+d*x]^j)^(m+j*k)*
        Sim[-b*A*n+a*(A+(A+C)*(j*k*m+(k+1)/2))*sin[c+d*x]^k+
          b*(A*(n+1)+(A+C)*(j*k*m+(k+1)/2))*sin[c+d*x]^(2*k),x]*
        (a+b*sin[c+d*x]^k)^(n-1),x]] /;
FreeQ[{a,b,c,d,A,C},x] && OneQ[j^2,k^2] && k2===2*k && NonzeroQ[a^2-b^2] &&
  RationalQ[m,n] && j*k*m+(k+1)/2≠0 && j*k*m≤-1 && n>0
```

- **Derivation: Recurrence 5**

- **Rule 5: If** $j^2 = k^2 = 1 \bigwedge a^2 - b^2 \neq 0 \bigwedge j\,k\,m + \frac{k+1}{2} \neq 0 \bigwedge j\,k\,m \leq -1 \bigwedge -1 \leq n < 0$, **then**

$$\int \left(\text{Sin}[c+d\,x]^j\right)^m \left(A + B\,\text{Sin}[c+d\,x]^k + C\,\text{Sin}[c+d\,x]^{2\,k}\right) \left(a + b\,\text{Sin}[c+d\,x]^k\right)^n dx \longrightarrow$$

$$\frac{A\,\text{Cos}[c+d\,x] \left(\text{Sin}[c+d\,x]^j\right)^{m+j\,k} \left(a+b\,\text{Sin}[c+d\,x]^k\right)^{n+1}}{a\,d\left(j\,k\,m+\frac{k+1}{2}\right)} + \frac{1}{a\left(j\,k\,m+\frac{k+1}{2}\right)} \int \left(\text{Sin}[c+d\,x]^j\right)^{m+j\,k} \cdot$$
$$\left((a\,B - b\,A)\left(j\,k\,m + \frac{k+1}{2}\right) - b\,A\,(n+1) + a\left(A + (A+C)\left(j\,k\,m+\frac{k+1}{2}\right)\right)\text{Sin}[c+d\,x]^k + \right.$$
$$\left. b\,A\left(j\,k\,m + n + \frac{k+5}{2}\right)\text{Sin}[c+d\,x]^{2\,k}\right) \left(a+b\,\text{Sin}[c+d\,x]^k\right)^n dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(A_.+B_.*sin[c_.+d_.*x_]^k_.+C_.*sin[c_.+d_.*x_]^k2_)*
    (a_+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  A*Cos[c+d*x]*(Sin[c+d*x]^j)^(m+j*k)*(a+b*Sin[c+d*x]^k)^(n+1)/(a*d*(j*k*m+(k+1)/2)) +
  Dist[1/(a*(j*k*m+(k+1)/2)),
    Int[(sin[c+d*x]^j)^(m+j*k)*
      Sim[(a*B-b*A)*(j*k*m+(k+1)/2)-b*A*(n+1)+a*(A+(A+C)*(j*k*m+(k+1)/2))*sin[c+d*x]^k+
        b*A*(j*k*m+n+(k+5)/2)*sin[c+d*x]^(2*k),x]*
      (a+b*sin[c+d*x]^k)^n,x]] /;
FreeQ[{a,b,c,d,A,B,C},x] && OneQ[j^2,k^2] && k2===2*k && NonzeroQ[a^2-b^2] &&
  RationalQ[m,n] && j*k*m+(k+1)/2≠0 && j*k*m≤-1 && -1≤n<0
```

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(A_.+C_.*sin[c_.+d_.*x_]^k2_)*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol
  A*Cos[c+d*x]*(Sin[c+d*x]^j)^(m+j*k)*(a+b*Sin[c+d*x]^k)^(n+1)/(a*d*(j*k*m+(k+1)/2)) +
  Dist[1/(a*(j*k*m+(k+1)/2)),
    Int[(sin[c+d*x]^j)^(m+j*k)*
      Sim[-b*A*(j*k*m+(k+1)/2)-b*A*(n+1)+a*(A+(A+C)*(j*k*m+(k+1)/2))*sin[c+d*x]^k+
        b*A*(j*k*m+n+(k+5)/2)*sin[c+d*x]^(2*k),x]*
      (a+b*sin[c+d*x]^k)^n,x]] /;
FreeQ[{a,b,c,d,A,C},x] && OneQ[j^2,k^2] && k2===2*k && NonzeroQ[a^2-b^2] &&
  RationalQ[m,n] && j*k*m+(k+1)/2≠0 && j*k*m≤-1 && -1≤n<0
```

- **Derivation: Recurrence 6**

- **Rule 6: If** $j^2 = k^2 = 1 \land a^2 - b^2 \neq 0 \land a^2 C - a b B + b^2 A \neq 0 \land j k m < 0 \land n < -1$, **then**

$$\int \left(\text{Sin}[c+d\,x]^j\right)^m \left(A + B\,\text{Sin}[c+d\,x]^k + C\,\text{Sin}[c+d\,x]^{2k}\right) \left(a + b\,\text{Sin}[c+d\,x]^k\right)^n \,dx \longrightarrow$$

$$\frac{\left(a^2 C - a b B + b^2 A\right) \text{Cos}[c+d\,x] \left(\text{Sin}[c+d\,x]^j\right)^{m+j k} \left(a + b\,\text{Sin}[c+d\,x]^k\right)^{n+1}}{a\,d\,(n+1)\left(a^2-b^2\right)} +$$

$$\frac{1}{a\,(n+1)\left(a^2-b^2\right)} \int \left(\text{Sin}[c+d\,x]^j\right)^m \cdot$$

$$\left(A\left(a^2-b^2\right)(n+1) - \left(a^2 C - a b B + b^2 A\right)\left(j k m + \frac{k+1}{2}\right) - a\,(b A - a B + b C)\,(n+1)\,\text{Sin}[c+d\,x]^k + \right.$$

$$\left.\left(a^2 C - a b B + b^2 A\right)\left(j k m + n + \frac{k+5}{2}\right)\text{Sin}[c+d\,x]^{2k}\right)\left(a + b\,\text{Sin}[c+d\,x]^k\right)^{n+1} dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(A_.+B_.*sin[c_.+d_.*x_]^k_.+C_.*sin[c_.+d_.*x_]^k2_)*
    (a_+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol] :=
  (a^2*C-a*b*B+b^2*A)*Cos[c+d*x]*(Sin[c+d*x]^j)^(m+j*k)*(a+b*Sin[c+d*x]^k)^(n+1)/(a*d*(n+1)*(a^2-b^2
  Dist[1/(a*(n+1)*(a^2-b^2)),
    Int[(sin[c+d*x]^j)^m*
      Sim[A*(a^2-b^2)*(n+1)-(a^2*C-a*b*B+b^2*A)*(j*k*m+(k+1)/2)-a*(b*A-a*B+b*C)*(n+1)*sin[c+d*x]^k+
        (a^2*C-a*b*B+b^2*A)*(j*k*m+n+(k+5)/2)*sin[c+d*x]^(2*k),x]*
       (a+b*sin[c+d*x]^k)^(n+1),x]] /;
FreeQ[{a,b,c,d,A,B,C},x] && OneQ[j^2,k^2] && k2===2*k && NonzeroQ[a^2-b^2] &&
  NonzeroQ[a^2*C-a*b*B+b^2*A] && RationalQ[m,n] && j*k*m<0 && n<-1
```

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(A_.+C_.*sin[c_.+d_.*x_]^k2_)*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol
  (a^2*C+b^2*A)*Cos[c+d*x]*(Sin[c+d*x]^j)^(m+j*k)*(a+b*Sin[c+d*x]^k)^(n+1)/(a*d*(n+1)*(a^2-b^2)) +
  Dist[1/(a*(n+1)*(a^2-b^2)),
    Int[(sin[c+d*x]^j)^m*
      Sim[A*(a^2-b^2)*(n+1)-(a^2*C+b^2*A)*(j*k*m+(k+1)/2)-a*b*(A+C)*(n+1)*sin[c+d*x]^k+
        (a^2*C+b^2*A)*(j*k*m+n+(k+5)/2)*sin[c+d*x]^(2*k),x]*
       (a+b*sin[c+d*x]^k)^(n+1),x]] /;
FreeQ[{a,b,c,d,A,C},x] && OneQ[j^2,k^2] && k2===2*k && NonzeroQ[a^2-b^2] &&
  NonzeroQ[a^2*C+b^2*A] && RationalQ[m,n] && j*k*m<0 && n<-1
```