$\int \text{Tanh}[a + b\,x]^n \, dx$

- **Reference: G&R 2.243.17, CRC 556, A&S 4.5.79**
- **Derivation: Reciprocal rule**
- **Basis:** $\text{Tanh}[z] = \frac{\text{Sinh}[z]}{\text{Cosh}[z]}$
- **Rule:**

$$\int \text{Tanh}[a + b\,x] \, dx \longrightarrow \frac{\text{Log}[\text{Cosh}[a + b\,x]]}{b}$$

- **Program code:**

```
Int[Tanh[a_.+b_.*x_],x_Symbol] :=
  Log[Cosh[a+b*x]]/b /;
FreeQ[{a,b},x]
```

- **Reference: G&R 2.423.33, CRC 557, A&S 4.5.82**

```
Int[Coth[a_.+b_.*x_],x_Symbol] :=
  Log[Sinh[a+b*x]]/b /;
FreeQ[{a,b},x]
```

- **Reference: G&R 2.423.22, CRC 569**
- **Derivation: Algebraic expansion**
- **Basis:** $\text{Tanh}[z]^2 = 1 - \text{Sech}[z]^2$
- **Rule:**

$$\int \text{Tanh}[a + b\,x]^2 \, dx \longrightarrow x - \frac{\text{Tanh}[a + b\,x]}{b}$$

- **Program code:**

```
Int[Tanh[a_.+b_.*x_]^2,x_Symbol] :=
  x - Tanh[a+b*x]/b /;
FreeQ[{a,b},x]
```

- **Reference: G&R 2.423.38, CRC 573**

```
Int[Coth[a_.+b_.*x_]^2,x_Symbol] :=
  x - Coth[a+b*x]/b /;
FreeQ[{a,b},x]
```

- **Reference: G&R 2.411.3, CRC 570, A&S 4.5.87**
- **Derivation: Integration by parts with a double-back flip**
- **Basis:** $\text{Tanh}[z]^n = \frac{\text{Tanh}[z]^{n-1}\,\text{Sinh}[z]}{\text{Cosh}[z]}$
- **Rule: If** $n > 1$**, then**

$$\int (c\,\text{Tanh}[a+b\,x])^n\,dx \longrightarrow -\frac{c\,(c\,\text{Tanh}[a+b\,x])^{n-1}}{b\,(n-1)} + c^2\int (c\,\text{Tanh}[a+b\,x])^{n-2}\,dx$$

- **Program code:**

```
Int[(c_.*Tanh[a_.+b_.*x_])^n_,x_Symbol] :=
  -c*(c*Tanh[a+b*x])^(n-1)/(b*(n-1)) +
  Dist[c^2,Int[(c*Tanh[a+b*x])^(n-2),x]] /;
FreeQ[{a,b,c},x] && RationalQ[n] && n>1
```

- **Reference: G&R 2.411.4, CRC 574, A&S 4.5.88**

```
Int[(c_.*Coth[a_.+b_.*x_])^n_,x_Symbol] :=
  -c*(c*Coth[a+b*x])^(n-1)/(b*(n-1)) +
  Dist[c^2,Int[(c*Coth[a+b*x])^(n-2),x]] /;
FreeQ[{a,b,c},x] && RationalQ[n] && n>1
```

- **Reference: G&R 2.411.4, CRC 574'**
- **Derivation: Inverted integration by parts with a double-back flip**
- **Rule: If** $n < -1$**, then**

$$\int (c\,\text{Tanh}[a+b\,x])^n\,dx \longrightarrow \frac{(c\,\text{Tanh}[a+b\,x])^{n+1}}{b\,c\,(n+1)} + \frac{1}{c^2}\int (c\,\text{Tanh}[a+b\,x])^{n+2}\,dx$$

- **Program code:**

```
Int[(c_.*Tanh[a_.+b_.*x_])^n_,x_Symbol] :=
  (c*Tanh[a+b*x])^(n+1)/(b*c*(n+1)) +
  Dist[1/c^2,Int[(c*Tanh[a+b*x])^(n+2),x]] /;
FreeQ[{a,b,c},x] && RationalQ[n] && n<-1
```

- **Reference: G&R 2.411.3, CRC 570'**

```
Int[(c_.*Coth[a_.+b_.*x_])^n_,x_Symbol] :=
  (c*Coth[a+b*x])^(n+1)/(b*c*(n+1)) +
  Dist[1/c^2,Int[(c*Coth[a+b*x])^(n+2),x]] /;
FreeQ[{a,b,c},x] && RationalQ[n] && n<-1
```

$\int (a + b \operatorname{Tanh}[c + d x])^n \, dx$ when $a^2 - b^2 = 0$

- **Rule: If** $a^2 - b^2 = 0$**, then**

$$\int \frac{1}{a + b \operatorname{Tanh}[c + d x]} \, dx \longrightarrow \frac{x}{2 a} - \frac{a}{2 b d (a + b \operatorname{Tanh}[c + d x])}$$

- **Program code:**

```
Int[1/(a_+b_.*Tanh[c_.+d_.*x_]),x_Symbol] :=
  x/(2*a) - a/(2*b*d*(a+b*Tanh[c+d*x])) /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2-b^2]
```

```
Int[1/(a_+b_.*Coth[c_.+d_.*x_]),x_Symbol] :=
  x/(2*a) - a/(2*b*d*(a+b*Coth[c+d*x])) /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2-b^2]
```

- **Rule: If** $a^2 - b^2 = 0 \land a > 0$**, then**

$$\int \sqrt{a + b \operatorname{Tanh}[c + d x]} \, dx \longrightarrow \frac{\sqrt{2}\, b}{d \sqrt{a}} \operatorname{ArcTanh}\left[\frac{\sqrt{a + b \operatorname{Tanh}[c + d x]}}{\sqrt{2} \sqrt{a}}\right]$$

- **Program code:**

```
Int[Sqrt[a_+b_.*Tanh[c_.+d_.*x_]],x_Symbol] :=
  Sqrt[2]*b/(d*Rt[a,2])*ArcTanh[Sqrt[a+b*Tanh[c+d*x]]/(Sqrt[2]*Rt[a,2])] /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2-b^2] && PosQ[a]
```

```
Int[Sqrt[a_+b_.*Coth[c_.+d_.*x_]],x_Symbol] :=
  (Sqrt[2]*b/(d*Rt[a,2])*ArcCoth[Sqrt[a+b*Coth[c+d*x]]/(Sqrt[2]*Rt[a,2])]) /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2-b^2] && PosQ[a]
```

- **Rule: If** $a^2 - b^2 = 0 \land \neg (a > 0)$**, then**

$$\int \sqrt{a + b \operatorname{Tanh}[c + d x]} \, dx \longrightarrow -\frac{\sqrt{2}\, b}{d \sqrt{-a}} \operatorname{ArcTan}\left[\frac{\sqrt{a + b \operatorname{Tanh}[c + d x]}}{\sqrt{2} \sqrt{-a}}\right]$$

- **Program code:**

```
Int[Sqrt[a_+b_.*Tanh[c_.+d_.*x_]],x_Symbol] :=
  -Sqrt[2]*b/(d*Rt[-a,2])*ArcTan[Sqrt[a+b*Tanh[c+d*x]]/(Sqrt[2]*Rt[-a,2])] /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2-b^2] && NegQ[a]
```

```
Int[Sqrt[a_+b_.*Coth[c_.+d_.*x_]],x_Symbol] :=
  Sqrt[2]*b/(d*Rt[-a,2])*ArcCot[Sqrt[a+b*Coth[c+d*x]]/(Sqrt[2]*Rt[-a,2])] /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2-b^2] && NegQ[a]
```

- **Rule: If** $a^2 - b^2 = 0 \wedge n \in \mathbb{F} \wedge n > 1$**, then**

$$\int (a + b \operatorname{Tanh}[c + d x])^n \, dx \longrightarrow -\frac{a^2 (a + b \operatorname{Tanh}[c + d x])^{n-1}}{b d (n-1)} + 2 a \int (a + b \operatorname{Tanh}[c + d x])^{n-1} \, dx$$

- **Program code:**

```
Int[(a_+b_.*Tanh[c_.+d_.*x_])^n_,x_Symbol] :=
  -a^2*(a+b*Tanh[c+d*x])^(n-1)/(b*d*(n-1)) +
  Dist[2*a,Int[(a+b*Tanh[c+d*x])^(n-1),x]] /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2-b^2] && FractionQ[n] && n>1
```

```
Int[(a_+b_.*Coth[c_.+d_.*x_])^n_,x_Symbol] :=
  -a^2*(a+b*Coth[c+d*x])^(n-1)/(b*d*(n-1)) +
  Dist[2*a,Int[(a+b*Coth[c+d*x])^(n-1),x]] /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2-b^2] && FractionQ[n] && n>1
```

- **Rule: If** $a^2 - b^2 = 0 \wedge n < 0$**, then**

$$\int (a + b \operatorname{Tanh}[c + d x])^n \, dx \longrightarrow \frac{a (a + b \operatorname{Tanh}[c + d x])^n}{2 b d n} + \frac{1}{2 a} \int (a + b \operatorname{Tanh}[c + d x])^{n+1} \, dx$$

- **Program code:**

```
Int[(a_+b_.*Tanh[c_.+d_.*x_])^n_,x_Symbol] :=
  a*(a+b*Tanh[c+d*x])^n/(2*b*d*n) +
  Dist[1/(2*a),Int[(a+b*Tanh[c+d*x])^(n+1),x]] /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2-b^2] && RationalQ[n] && n<0
```

```
Int[(a_+b_.*Coth[c_.+d_.*x_])^n_,x_Symbol] :=
  a*(a+b*Coth[c+d*x])^n/(2*b*d*n) +
  Dist[1/(2*a),Int[(a+b*Coth[c+d*x])^(n+1),x]] /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2-b^2] && RationalQ[n] && n<0
```

$\int (a + b\,\mathrm{Tanh}[c + d\,x])^n \, dx$ when $a^2 + b^2 \neq 0$

- **Derivation: Algebraic expansion and integration by substitution**

- **Basis:** $\frac{1}{a+b\,\mathrm{Tanh}[z]} = \frac{a}{a^2-b^2} - \frac{b}{(a^2-b^2)\,(a\,\mathrm{Cosh}[z]+b\,\mathrm{Sinh}[z])}\,\partial_z\,(a\,\mathrm{Cosh}[z] + b\,\mathrm{Sinh}[z])$

- **Rule: If** $a^2 - b^2 \neq 0$**, then**

$$\int \frac{1}{a + b\,\mathrm{Tanh}[c + d\,x]} \, dx \longrightarrow \frac{a\,x}{a^2 - b^2} - \frac{b\,\mathrm{Log}[a\,\mathrm{Cosh}[c + d\,x] + b\,\mathrm{Sinh}[c + d\,x]]}{d\,(a^2 - b^2)}$$

- **Program code:**

```
Int[1/(a_+b_.*Tanh[c_.+d_.*x_]),x_Symbol] :=
  a*x/(a^2-b^2) - b*Log[a*Cosh[c+d*x]+b*Sinh[c+d*x]]/(d*(a^2-b^2)) /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2-b^2]
```

```
Int[1/(a_+b_.*Coth[c_.+d_.*x_]),x_Symbol] :=
  a*x/(a^2-b^2) - b*Log[b*Cosh[c+d*x]+a*Sinh[c+d*x]]/(d*(a^2-b^2)) /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2-b^2]
```

$$\int (A + B\,\mathrm{Tanh}[c + d\,x])\,(a + b\,\mathrm{Tanh}[c + d\,x])^n\,dx$$

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{A+B z}{a+b z} = \frac{B}{b} + \frac{b A-a B}{b}\,\frac{1}{a+b z}$
- **Rule: If** $b A - a B \neq 0$, **then**

$$\int \frac{A + B\,\mathrm{Tanh}[c + d\,x]}{a + b\,\mathrm{Tanh}[c + d\,x]}\,dx \longrightarrow \frac{B\,x}{b} + \frac{b A - a B}{b}\int \frac{1}{a + b\,\mathrm{Tanh}[c + d\,x]}\,dx$$

- **Program code:**

```
Int[(A_.+B_.*Tanh[c_.+d_.*x_])/(a_.+b_.*Tanh[c_.+d_.*x_]),x_Symbol] :=
  B*x/b + Dist[(b*A-a*B)/b,Int[1/(a+b*Tanh[c+d*x]),x]] /;
FreeQ[{a,b,c,d,A,B},x] && NonzeroQ[b*A-a*B]
```

```
Int[(A_.+B_.*Coth[c_.+d_.*x_])/(a_.+b_.*Coth[c_.+d_.*x_]),x_Symbol] :=
  B*x/b + Dist[(b*A-a*B)/b,Int[1/(a+b*Coth[c+d*x]),x]] /;
FreeQ[{a,b,c,d,A,B},x] && NonzeroQ[b*A-a*B]
```

- **Note: This rule does not appear in published integral tables.**
- **Rule: If** $A^2 - B^2 = 0 \land b A + a B \neq 0$, **then**

$$\int \frac{A + B\,\mathrm{Tanh}[c + d\,x]}{\sqrt{a + b\,\mathrm{Tanh}[c + d\,x]}}\,dx \longrightarrow \frac{2 B}{d\sqrt{\frac{b A+a B}{B}}}\,\mathrm{ArcTanh}\left[\frac{\sqrt{a + b\,\mathrm{Tanh}[c + d\,x]}}{\sqrt{\frac{b A+a B}{B}}}\right]$$

- **Program code:**

```
Int[(A_+B_.*Tanh[c_.+d_.*x_])/Sqrt[a_.+b_.*Tanh[c_.+d_.*x_]],x_Symbol] :=
  2*B/(d*Sqrt[(b*A+a*B)/B])*ArcTanh[Sqrt[a+b*Tanh[c+d*x]]/Sqrt[(b*A+a*B)/B]] /;
FreeQ[{a,b,c,d,A,B},x] && ZeroQ[A^2-B^2] && NonzeroQ[b*A+a*B]
```

```
Int[(A_+B_.*Coth[c_.+d_.*x_])/Sqrt[a_.+b_.*Coth[c_.+d_.*x_]],x_Symbol] :=
  2*B/(d*Sqrt[(b*A+a*B)/B])*ArcTanh[Sqrt[a+b*Coth[c+d*x]]/Sqrt[(b*A+a*B)/B]] /;
FreeQ[{a,b,c,d,A,B},x] && ZeroQ[A^2-B^2] && NonzeroQ[b*A+a*B]
```

- **Derivation: Algebraic expansion**
- **Basis:** $A + B z = \frac{A+B}{2} (1 + z) + \frac{A-B}{2} (1 - z)$
- **Rule: If** $A^2 - B^2 \neq 0 \land a^2 - b^2 \neq 0$, **then**

$$\int \frac{A + B \operatorname{Tanh}[c + d x]}{\sqrt{a + b \operatorname{Tanh}[c + d x]}} \, dx \longrightarrow \frac{A + B}{2} \int \frac{1 + \operatorname{Tanh}[c + d x]}{\sqrt{a + b \operatorname{Tanh}[c + d x]}} \, dx + \frac{A - B}{2} \int \frac{1 - \operatorname{Tanh}[c + d x]}{\sqrt{a + b \operatorname{Tanh}[c + d x]}} \, dx$$

- **Program code:**

```
Int[(A_.+B_.*Tanh[c_.+d_.*x_])/Sqrt[a_.+b_.*Tanh[c_.+d_.*x_]],x_Symbol] :=
  Dist[(A+B)/2,Int[(1+Tanh[c+d*x])/Sqrt[a+b*Tanh[c+d*x]],x]] +
  Dist[(A-B)/2,Int[(1-Tanh[c+d*x])/Sqrt[a+b*Tanh[c+d*x]],x]] /;
FreeQ[{a,b,c,d,A,B},x] && NonzeroQ[A^2-B^2] && NonzeroQ[a^2-b^2]
```

```
Int[(A_.+B_.*Coth[c_.+d_.*x_])/Sqrt[a_.+b_.*Coth[c_.+d_.*x_]],x_Symbol] :=
  Dist[(A+B)/2,Int[(1+Coth[c+d*x])/Sqrt[a+b*Coth[c+d*x]],x]] +
  Dist[(A-B)/2,Int[(1-Coth[c+d*x])/Sqrt[a+b*Coth[c+d*x]],x]] /;
FreeQ[{a,b,c,d,A,B},x] && NonzeroQ[A^2-B^2] && NonzeroQ[a^2-b^2]
```

- **Note: This rule does not appear in published integral tables.**
- **Rule: If** $n > 0$, **then**

$$\int (A + B \operatorname{Tanh}[c + d x]) (a + b \operatorname{Tanh}[c + d x])^n \, dx \longrightarrow$$
$$-\frac{B (a + b \operatorname{Tanh}[c + d x])^n}{d n} + \int (a A + b B + (b A + a B) \operatorname{Tanh}[c + d x]) (a + b \operatorname{Tanh}[c + d x])^{n-1} \, dx$$

- **Program code:**

```
Int[(A_.+B_.*Tanh[c_.+d_.*x_])*(a_+b_.*Tanh[c_.+d_.*x_])^n_.,x_Symbol] :=
  -B*(a+b*Tanh[c+d*x])^n/(d*n) +
  Int[(a*A+b*B+(b*A+a*B)*Tanh[c+d*x])*(a+b*Tanh[c+d*x])^(n-1),x] /;
FreeQ[{a,b,c,d,A,B},x] && RationalQ[n] && n>0
```

```
Int[(A_.+B_.*Coth[c_.+d_.*x_])*(a_+b_.*Coth[c_.+d_.*x_])^n_.,x_Symbol] :=
  -B*(a+b*Coth[c+d*x])^n/(d*n) +
  Int[(a*A+b*B+(b*A+a*B)*Coth[c+d*x])*(a+b*Coth[c+d*x])^(n-1),x] /;
FreeQ[{a,b,c,d,A,B},x] && RationalQ[n] && n>0
```

- **Note: This rule does not appear in published integral tables.**

- **Rule: If** $a^2 - b^2 \neq 0 \wedge n < -1$, **then**

$$\int (A + B \operatorname{Tanh}[c + d x]) (a + b \operatorname{Tanh}[c + d x])^n \, dx \longrightarrow$$

$$-\frac{(b A - a B) (a + b \operatorname{Tanh}[c + d x])^{n+1}}{d \left(a^2 - b^2\right) (n + 1)} +$$

$$\frac{1}{a^2 - b^2} \int (a A - b B - (b A - a B) \operatorname{Tanh}[c + d x]) (a + b \operatorname{Tanh}[c + d x])^{n+1} \, dx$$

- **Program code:**

```
Int[(A_.+B_.*Tanh[c_.+d_.*x_])*(a_+b_.*Tanh[c_.+d_.*x_])^n_,x_Symbol] :=
  -(b*A-a*B)*(a+b*Tanh[c+d*x])^(n+1)/(d*(a^2-b^2)*(n+1)) +
  Dist[1/(a^2-b^2),Int[(a*A-b*B-(b*A-a*B)*Tanh[c+d*x])*(a+b*Tanh[c+d*x])^(n+1),x]] /;
FreeQ[{a,b,c,d,A,B},x] && NonzeroQ[a^2-b^2] && RationalQ[n] && n<-1
```

```
Int[(A_.+B_.*Coth[c_.+d_.*x_])*(a_+b_.*Coth[c_.+d_.*x_])^n_,x_Symbol] :=
  -(b*A-a*B)*(a+b*Coth[c+d*x])^(n+1)/(d*(a^2-b^2)*(n+1)) +
  Dist[1/(a^2-b^2),Int[(a*A-b*B-(b*A-a*B)*Coth[c+d*x])*(a+b*Coth[c+d*x])^(n+1),x]] /;
FreeQ[{a,b,c,d,A,B},x] && NonzeroQ[a^2-b^2] && RationalQ[n] && n<-1
```

$$\int \left(a + b\,\mathrm{Tan}[c + d\,x]^2\right)^n \,dx$$

- **Derivation: Algebraic simplification**
- **Basis: If** $a - b = 0$**, then** $a + b\,\mathrm{Tan}[z]^2 = b\,\mathrm{Sec}[z]^2$
- **Rule: If** $a - b = 0 \bigwedge m \in \mathbb{Z}$**, then**

$$\int u\,\left(a + b\,\mathrm{Tan}[v]^2\right)^m dx \;\longrightarrow\; b^m \int u\,\mathrm{Sec}[v]^{2m}\,dx$$

- **Program code:**

```
Int[u_.*(a_+b_.*Tan[v_]^2)^m_,x_Symbol] :=
  Dist[b^m,Int[u*Sec[v]^(2*m),x]] /;
FreeQ[{a,b,m},x] && ZeroQ[a-b] && IntegerQ[m]
```

- **Derivation: Algebraic simplification**
- **Basis: If** $a - b = 0$**, then** $a + b\,\mathrm{Tan}[z]^2 = b\,\mathrm{Sec}[z]^2$
- **Rule: If** $a - b = 0 \bigwedge m \notin \mathbb{Z}$**, then**

$$\int u\,\left(a + b\,\mathrm{Tan}[v]^2\right)^m dx \;\longrightarrow\; \int u\,\left(b\,\mathrm{Sec}[v]^2\right)^m dx$$

- **Program code:**

```
Int[u_.*(a_+b_.*Tan[v_]^2)^m_,x_Symbol] :=
  Int[u*(b*Sec[v]^2)^m,x] /;
FreeQ[{a,b,m},x] && ZeroQ[a-b] && Not[IntegerQ[m]]
```

- **Rule: If** $a + b \neq 0$**, then**

$$\int \frac{1}{a + b\,\mathrm{Tanh}[c + d\,x]^2}\,dx \;\longrightarrow\; \frac{x}{a + b} + \frac{\sqrt{b}}{\sqrt{a}\;d\,(a + b)}\,\mathrm{ArcTan}\left[\frac{\sqrt{b}\;\mathrm{Tanh}[c + d\,x]}{\sqrt{a}}\right]$$

- **Program code:**

```
Int[1/(a_+b_.*Tanh[c_.+d_.*x_]^2),x_Symbol] :=
  x/(a+b) + Sqrt[b]*ArcTan[Sqrt[b]*Tanh[c+d*x]/Sqrt[a]]/(Sqrt[a]*d*(a+b)) /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a+b]
```

```
Int[1/(a_+b_.*Coth[c_.+d_.*x_]^2),x_Symbol] :=
  x/(a+b) + Sqrt[b]*ArcTan[Sqrt[b]*Coth[c+d*x]/Sqrt[a]]/(Sqrt[a]*d*(a+b)) /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a+b]
```

$$\int x^m \operatorname{Tanh}[a+b\,x^n]^p\,dx$$

- **Derivation: Algebraic expansion**
- **Basis:** $\operatorname{Tanh}[z]=1-\frac{2}{1+e^{2z}}$
- **Rule: If** $m\in\mathbb{Z}\wedge m>0\wedge m-n+1\neq 0$**, then**

$$\int x^m \operatorname{Tanh}[a+b\,x^n]\,dx\;\longrightarrow\;\frac{x^{m+1}}{m+1}-2\int\frac{x^m}{1+e^{2a+2b\,x^n}}\,dx$$

- **Program code:**

```
Int[x_^m_.*Tanh[a_.+b_.*x_^n_.],x_Symbol] :=
  x^(m+1)/(m+1) -
  Dist[2,Int[x^m/(1+E^(2*a+2*b*x^n)),x]] /;
FreeQ[{a,b,n},x] && IntegerQ[m] && m>0 && NonzeroQ[m-n+1]
```

```
Int[x_^m_.*Coth[a_.+b_.*x_^n_.],x_Symbol] :=
  x^(m+1)/(m+1) -
  Dist[2,Int[x^m/(1-E^(2*a+2*b*x^n)),x]] /;
FreeQ[{a,b,n},x] && IntegerQ[m] && m>0 && NonzeroQ[m-n+1]
```

- **Note: This rule does not appear in published integral tables.**
- **Rule: If** $p>1\wedge m-n+1\neq 0\wedge 0<n\leq m$**, then**

$$\int x^m \operatorname{Tanh}[a+b\,x^n]^p\,dx\;\longrightarrow\;-\frac{x^{m-n+1}\operatorname{Tanh}[a+b\,x^n]^{p-1}}{b\,n\,(p-1)}+\frac{m-n+1}{b\,n\,(p-1)}\int x^{m-n}\operatorname{Tanh}[a+b\,x^n]^{p-1}\,dx+\int x^m\operatorname{Tanh}[a+b\,x^n]^{p-2}\,dx$$

- **Program code:**

```
Int[x_^m_.*Tanh[a_.+b_.*x_^n_.]^p_,x_Symbol] :=
  -x^(m-n+1)*Tanh[a+b*x^n]^(p-1)/(b*n*(p-1)) +
  Dist[(m-n+1)/(b*n*(p-1)),Int[x^(m-n)*Tanh[a+b*x^n]^(p-1),x]] +
  Int[x^m*Tanh[a+b*x^n]^(p-2),x] /;
FreeQ[{a,b},x] && RationalQ[{m,n,p}] && p>1 && NonzeroQ[m-n+1] && 0<n<=m
```

```
Int[x_^m_.*Coth[a_.+b_.*x_^n_.]^p_,x_Symbol] :=
  -x^(m-n+1)*Coth[a+b*x^n]^(p-1)/(b*n*(p-1)) +
  Dist[(m-n+1)/(b*n*(p-1)),Int[x^(m-n)*Coth[a+b*x^n]^(p-1),x]] +
  Int[x^m*Coth[a+b*x^n]^(p-2),x] /;
FreeQ[{a,b},x] && RationalQ[{m,n,p}] && p>1 && NonzeroQ[m-n+1] && 0<n<=m
```

$\int x^m \operatorname{Tanh}[a + b x + c x^2] \, dx$

- **Rule:**

$$\int x \operatorname{Tanh}[a + b x + c x^2] \, dx \longrightarrow \frac{\operatorname{Log}[\operatorname{Cosh}[a + b x + c x^2]]}{2 c} - \frac{b}{2 c} \int \operatorname{Tanh}[a + b x + c x^2] \, dx$$

- **Program code:**

```
Int[x_*Tanh[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  Log[Cosh[a+b*x+c*x^2]]/(2*c) -
  Dist[b/(2*c),Int[Tanh[a+b*x+c*x^2],x]] /;
FreeQ[{a,b,c},x]
```

```
Int[x_*Coth[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  Log[Sinh[a+b*x+c*x^2]]/(2*c) -
  Dist[b/(2*c),Int[Coth[a+b*x+c*x^2],x]] /;
FreeQ[{a,b,c},x]
```

- **Note: This rule is valid, but to be useful need a rule for reducing integrands of the form $x^m \operatorname{Log}[\operatorname{Cosh}[a + b x + c x^2]]$.**

- **Rule: If $m > 1$, then**

$$\int x^m \operatorname{Tanh}[a + b x + c x^2] \, dx \longrightarrow \frac{x^{m-1} \operatorname{Log}[\operatorname{Cosh}[a + b x + c x^2]]}{2 c} - \frac{b}{2 c} \int x^{m-1} \operatorname{Tanh}[a + b x + c x^2] \, dx - \frac{m-1}{2 c} \int x^{m-2} \operatorname{Log}[\operatorname{Cosh}[a + b x + c x^2]] \, dx$$

- **Program code:**

```
(* Int[x_^m_*Tanh[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  x^(m-1)*Log[Cosh[a+b*x+c*x^2]]/(2*c) -
  Dist[b/(2*c),Int[x^(m-1)*Tanh[a+b*x+c*x^2],x]] -
  Dist[(m-1)/(2*c),Int[x^(m-2)*Log[Cosh[a+b*x+c*x^2]],x]] /;
FreeQ[{a,b,c},x] && RationalQ[m] && m>1 *)
```

```
(* Int[x_^m_*Coth[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  x^(m-1)*Log[Sinh[a+b*x+c*x^2]]/(2*c) -
  Dist[b/(2*c),Int[x^(m-1)*Coth[a+b*x+c*x^2],x]] -
  Dist[(m-1)/(2*c),Int[x^(m-2)*Log[Sinh[a+b*x+c*x^2]],x]] /;
FreeQ[{a,b,c},x] && RationalQ[m] && m>1 *)
```