$$\int \frac{x\,(A+B\,\mathrm{Sin}[c+d\,x])}{(a+b\,\mathrm{Sin}[c+d\,x])^2}\,dx$$

- **Derivation: Integration by parts**
- **Rule: If** $a\,A-b\,B=0$**, then**

$$\int \frac{x\,(A+B\,\mathrm{Sin}[c+d\,x])}{(a+b\,\mathrm{Sin}[c+d\,x])^2}\,dx \longrightarrow -\frac{B\,x\,\mathrm{Cos}[c+d\,x]}{a\,d\,(a+b\,\mathrm{Sin}[c+d\,x])}+\frac{B}{a\,d}\int \frac{\mathrm{Cos}[c+d\,x]}{a+b\,\mathrm{Sin}[c+d\,x]}\,dx$$

- **Program code:**

```
Int[x_*(A_+B_.*Sin[c_.+d_.*x_])/(a_+b_.*Sin[c_.+d_.*x_])^2,x_Symbol] :=
  -B*x*Cos[c+d*x]/(a*d*(a+b*Sin[c+d*x])) +
  Dist[B/(a*d),Int[Cos[c+d*x]/(a+b*Sin[c+d*x]),x]] /;
FreeQ[{a,b,c,d,A,B},x] && ZeroQ[a*A-b*B]
```

```
Int[x_*(A_+B_.*Cos[c_.+d_.*x_])/(a_+b_.*Cos[c_.+d_.*x_])^2,x_Symbol] :=
  B*x*Sin[c+d*x]/(a*d*(a+b*Cos[c+d*x])) -
  Dist[B/(a*d),Int[Sin[c+d*x]/(a+b*Cos[c+d*x]),x]] /;
FreeQ[{a,b,c,d,A,B},x] && ZeroQ[a*A-b*B]
```

$$\int \mathrm{Sin}[a+b\,x]^m\,\mathrm{Tan}[a+b\,x]^n\,dx$$

- **Reference: G&R 2.526.18', CRC 327'**
- **Derivation: Algebraic expansion**
- **Basis:** $\mathrm{Sin}[z]\,\mathrm{Tan}[z] = -\mathrm{Cos}[z] + \mathrm{Sec}[z]$
- **Rule:**

$$\int \mathrm{Sin}[a+b\,x]\,\mathrm{Tan}[a+b\,x]\,dx \longrightarrow -\frac{\mathrm{Sin}[a+b\,x]}{b} + \int \mathrm{Sec}[a+b\,x]\,dx$$

- **Program code:**

```
Int[Sin[a_.+b_.*x_]*Tan[a_.+b_.*x_],x_Symbol] :=
  -Sin[a+b*x]/b + Int[Sec[a+b*x],x] /;
FreeQ[{a,b},x]
```

- **Reference: G&R 2.526.34'**

```
Int[Cos[a_.+b_.*x_]*Cot[a_.+b_.*x_],x_Symbol] :=
  Cos[a+b*x]/b + Int[Csc[a+b*x],x] /;
FreeQ[{a,b},x]
```

- **Rule: If** $m+n-1=0$**, then**

$$\int \mathrm{Sin}[a+b\,x]^m\,\mathrm{Tan}[a+b\,x]^n\,dx \longrightarrow -\frac{\mathrm{Sin}[a+b\,x]^m\,\mathrm{Tan}[a+b\,x]^{n-1}}{b\,m}$$

- **Program code:**

```
Int[Sin[a_.+b_.*x_]^m_*Tan[a_.+b_.*x_]^n_,x_Symbol] :=
  -Sin[a+b*x]^m*Tan[a+b*x]^(n-1)/(b*m) /;
FreeQ[{a,b,m,n},x] && ZeroQ[m+n-1]
```

```
Int[Cos[a_.+b_.*x_]^m_*Cot[a_.+b_.*x_]^n_,x_Symbol] :=
  Cos[a+b*x]^m*Cot[a+b*x]^(n-1)/(b*m) /;
FreeQ[{a,b,m,n},x] && ZeroQ[m+n-1]
```

- **Derivation: Integration by substitution**

- **Basis: If** $m, n, \frac{m+n-1}{2} \in \mathbb{Z}$**, then** $\text{Sin}[z]^m\, \text{Tan}[z]^n = -\frac{\left(1-\text{Cos}[z]^2\right)^{\frac{m+n-1}{2}}}{\text{Cos}[z]^n}\, \partial_z \text{Cos}[z]$

- **Note: This rule is used if** $m + n$ **is odd since it requires fewer steps and results in a simpler antiderivative than the other rules in this section.**

- **Rule: If** $m, n, \frac{m+n-1}{2} \in \mathbb{Z}$**, then**

$$\int \text{Sin}[a+b\,x]^m\, \text{Tan}[a+b\,x]^n\, dx \longrightarrow -\frac{1}{b}\, \text{Subst}\left[\int \frac{\left(1-x^2\right)^{\frac{m+n-1}{2}}}{x^n}\, dx,\ x,\ \text{Cos}[a+b\,x]\right]$$

- **Program code:**

```
Int[Sin[a_.+b_.*x_]^m_.*Tan[a_.+b_.*x_]^n_.,x_Symbol] :=
  Dist[-1/b,Subst[Int[Regularize[(1-x^2)^((m+n-1)/2)/x^n,x],x],x,Cos[a+b*x]]] /;
FreeQ[{a,b},x] && IntegersQ[m,n,(m+n-1)/2]
```

- **Basis: If** $m, n, \frac{m+n-1}{2} \in \mathbb{Z}$**, then** $\text{Cos}[z]^m\, \text{Cot}[z]^n = \frac{\left(1-\text{Sin}[z]^2\right)^{\frac{m+n-1}{2}}}{\text{Sin}[z]^n}\, \partial_z \text{Sin}[z]$

```
Int[Cos[a_.+b_.*x_]^m_.*Cot[a_.+b_.*x_]^n_.,x_Symbol] :=
  Dist[1/b,Subst[Int[Regularize[(1-x^2)^((m+n-1)/2)/x^n,x],x],x,Sin[a+b*x]]] /;
FreeQ[{a,b},x] && IntegersQ[m,n,(m+n-1)/2]
```

- **Reference: G&R 2.510.5, CRC 323a**

- **Rule: If** $m > 1 \land n < -1$**, then**

$$\int \text{Sin}[a+b\,x]^m\, \text{Tan}[a+b\,x]^n\, dx \longrightarrow \frac{\text{Sin}[a+b\,x]^m\, \text{Tan}[a+b\,x]^{n+1}}{b\,m} - \frac{n+1}{m} \int \text{Sin}[a+b\,x]^{m-2}\, \text{Tan}[a+b\,x]^{n+2}\, dx$$

- **Program code:**

```
Int[Sin[a_.+b_.*x_]^m_*Tan[a_.+b_.*x_]^n_,x_Symbol] :=
  Sin[a+b*x]^m*Tan[a+b*x]^(n+1)/(b*m) -
  Dist[(n+1)/m,Int[Sin[a+b*x]^(m-2)*Tan[a+b*x]^(n+2),x]] /;
FreeQ[{a,b},x] && RationalQ[{m,n}] && m>1 && n<-1
```

- **Reference: G&R 2.510.2, CRC 323b**

```
Int[Cos[a_.+b_.*x_]^m_*Cot[a_.+b_.*x_]^n_,x_Symbol] :=
  -Cos[a+b*x]^m*Cot[a+b*x]^(n+1)/(b*m) -
  Dist[(n+1)/m,Int[Cos[a+b*x]^(m-2)*Cot[a+b*x]^(n+2),x]] /;
FreeQ[{a,b},x] && RationalQ[{m,n}] && m>1 && n<-1
```

- **Reference: G&R 2.510.6, CRC 334b**
- **Rule: If $m < -1 \wedge n > 1$, then**

$$\int \mathrm{Sin}[a+b\,x]^m\,\mathrm{Tan}[a+b\,x]^n\,dx \longrightarrow \frac{\mathrm{Sin}[a+b\,x]^{m+2}\,\mathrm{Tan}[a+b\,x]^{n-1}}{b\,(n-1)} - \frac{m+2}{n-1}\int \mathrm{Sin}[a+b\,x]^{m+2}\,\mathrm{Tan}[a+b\,x]^{n-2}\,dx$$

- **Program code:**

```
Int[Sin[a_.+b_.*x_]^m_*Tan[a_.+b_.*x_]^n_,x_Symbol] :=
  Sin[a+b*x]^(m+2)*Tan[a+b*x]^(n-1)/(b*(n-1)) -
  Dist[(m+2)/(n-1),Int[Sin[a+b*x]^(m+2)*Tan[a+b*x]^(n-2),x]] /;
FreeQ[{a,b},x] && RationalQ[{m,n}] && m<-1 && n>1
```

- **Reference: G&R 2.510.3, CRC 334a**

```
Int[Cos[a_.+b_.*x_]^m_*Cot[a_.+b_.*x_]^n_,x_Symbol] :=
  -Cos[a+b*x]^(m+2)*Cot[a+b*x]^(n-1)/(b*(n-1)) -
  Dist[(m+2)/(n-1),Int[Cos[a+b*x]^(m+2)*Cot[a+b*x]^(n-2),x]] /;
FreeQ[{a,b},x] && RationalQ[{m,n}] && m<-1 && n>1
```

- **Reference: G&R 2.510.2, CRC 323b**
- **Rule: If $m > 1$, then**

$$\int \mathrm{Sin}[a+b\,x]^m\,\mathrm{Tan}[a+b\,x]^n\,dx \longrightarrow -\frac{\mathrm{Sin}[a+b\,x]^{m}\,\mathrm{Tan}[a+b\,x]^{n-1}}{b\,m} + \frac{m+n-1}{m}\int \mathrm{Sin}[a+b\,x]^{m-2}\,\mathrm{Tan}[a+b\,x]^{n}\,dx$$

- **Program code:**

```
Int[Sin[a_.+b_.*x_]^m_*Tan[a_.+b_.*x_]^n_.,x_Symbol]:=
  -Sin[a+b*x]^m*Tan[a+b*x]^(n-1)/(b*m) +
  Dist[(m+n-1)/m,Int[Sin[a+b*x]^(m-2)*Tan[a+b*x]^n,x]] /;
FreeQ[{a,b,n},x] && RationalQ[m] && m>1
```

- **Reference: G&R 2.510.5, CRC 323a**

```
Int[Cos[a_.+b_.*x_]^m_*Cot[a_.+b_.*x_]^n_.,x_Symbol] :=
  Cos[a+b*x]^m*Cot[a+b*x]^(n-1)/(b*m) +
  Dist[(m+n-1)/m,Int[Cos[a+b*x]^(m-2)*Cot[a+b*x]^n,x]] /;
FreeQ[{a,b,n},x] && RationalQ[m] && m>1
```

- **Reference: G&R 2.510.1**
- **Rule: If $n > 1$, then**

$$\int \text{Sin}[a+b\,x]^m\,\text{Tan}[a+b\,x]^n\,dx \longrightarrow \frac{\text{Sin}[a+b\,x]^m\,\text{Tan}[a+b\,x]^{n-1}}{b\,(n-1)} - \frac{m+n-1}{n-1}\int \text{Sin}[a+b\,x]^m\,\text{Tan}[a+b\,x]^{n-2}\,dx$$

- **Program code:**

```
Int[Sin[a_.+b_.*x_]^m_.*Tan[a_.+b_.*x_]^n_,x_Symbol] :=
  Sin[a+b*x]^m*Tan[a+b*x]^(n-1)/(b*(n-1)) -
  Dist[(m+n-1)/(n-1),Int[Sin[a+b*x]^m*Tan[a+b*x]^(n-2),x]] /;
FreeQ[{a,b,m},x] && RationalQ[n] && n>1
```

- **Reference: G&R 2.510.4**

```
Int[Cos[a_.+b_.*x_]^m_.*Cot[a_.+b_.*x_]^n_,x_Symbol] :=
  -Cos[a+b*x]^m*Cot[a+b*x]^(n-1)/(b*(n-1)) -
  Dist[(m+n-1)/(n-1),Int[Cos[a+b*x]^m*Cot[a+b*x]^(n-2),x]] /;
FreeQ[{a,b,m},x] && RationalQ[n] && n>1
```

- **Reference: G&R 2.510.3, CRC 334a**
- **Rule: If $m < -1 \land m+n+1 \neq 0$, then**

$$\int \text{Sin}[a+b\,x]^m\,\text{Tan}[a+b\,x]^n\,dx \longrightarrow \frac{\text{Sin}[a+b\,x]^{m+2}\,\text{Tan}[a+b\,x]^{n-1}}{b\,(m+n+1)} + \frac{m+2}{m+n+1}\int \text{Sin}[a+b\,x]^{m+2}\,\text{Tan}[a+b\,x]^{n}\,dx$$

- **Program code:**

```
Int[Sin[a_.+b_.*x_]^m_*Tan[a_.+b_.*x_]^n_.,x_Symbol]:=
  Sin[a+b*x]^(m+2)*Tan[a+b*x]^(n-1)/(b*(m+n+1)) +
  Dist[(m+2)/(m+n+1),Int[Sin[a+b*x]^(m+2)*Tan[a+b*x]^n,x]] /;
FreeQ[{a,b,n},x] && RationalQ[m] && m<-1 && NonzeroQ[m+n+1]
```

- **Reference: G&R 2.510.6, CRC 334b**

```
Int[Cos[a_.+b_.*x_]^m_*Cot[a_.+b_.*x_]^n_.,x_Symbol] :=
  -Cos[a+b*x]^(m+2)*Cot[a+b*x]^(n-1)/(b*(m+n+1)) +
  Dist[(m+2)/(m+n+1),Int[Cos[a+b*x]^(m+2)*Cot[a+b*x]^n,x]] /;
FreeQ[{a,b,n},x] && RationalQ[m] && m<-1 && NonzeroQ[m+n+1]
```

- **Reference: G&R 2.510.4**

- **Rule: If** $n < -1 \land m+n+1 \neq 0$**, then**

$$\int \mathrm{Sin}[a+b\,x]^m\,\mathrm{Tan}[a+b\,x]^n\,dx \longrightarrow \frac{\mathrm{Sin}[a+b\,x]^m\,\mathrm{Tan}[a+b\,x]^{n+1}}{b\,(m+n+1)} - \frac{n+1}{m+n+1}\int \mathrm{Sin}[a+b\,x]^m\,\mathrm{Tan}[a+b\,x]^{n+2}\,dx$$

- **Program code:**

```
Int[Sin[a_.+b_.*x_]^m_.*Tan[a_.+b_.*x_]^n_,x_Symbol]:=
  Sin[a+b*x]^m*Tan[a+b*x]^(n+1)/(b*(m+n+1)) -
  Dist[(n+1)/(m+n+1),Int[Sin[a+b*x]^m*Tan[a+b*x]^(n+2),x]] /;
FreeQ[{a,b,m},x] && RationalQ[n] && n<-1 && NonzeroQ[m+n+1]
```

- **Reference: G&R 2.510.1**

```
Int[Cos[a_.+b_.*x_]^m_.*Cot[a_.+b_.*x_]^n_,x_Symbol] :=
  -Cos[a+b*x]^m*Cot[a+b*x]^(n+1)/(b*(m+n+1)) -
  Dist[(n+1)/(m+n+1),Int[Cos[a+b*x]^m*Cot[a+b*x]^(n+2),x]] /;
FreeQ[{a,b,m},x] && RationalQ[n] && n<-1 && NonzeroQ[m+n+1]
```

$$\int \mathrm{Sin}[a+b\,x]\ \mathrm{Sin}[2\,(a+b\,x)]^n\,dx$$

- **Rule:**

$$\int \frac{\mathrm{Sin}[a+b\,x]}{\sqrt{\mathrm{Sin}[2\,(a+b\,x)]}}\,dx \longrightarrow -\frac{\mathrm{ArcSin}[\mathrm{Cos}[a+b\,x]-\mathrm{Sin}[a+b\,x]]}{2\,b}-\frac{\mathrm{Log}\Big[\mathrm{Cos}[a+b\,x]+\mathrm{Sin}[a+b\,x]+\sqrt{\mathrm{Sin}[2\,(a+b\,x)]}\Big]}{2\,b}$$

- **Program code:**

```
Int[Sin[a_.+b_.*x_]/Sqrt[Sin[c_.+d_.*x_]],x_Symbol] :=
  -ArcSin[Cos[a+b*x]-Sin[a+b*x]]/(2*b) - Log[Cos[a+b*x]+Sin[a+b*x]+Sqrt[Sin[c+d*x]]]/(2*b) /;
FreeQ[{a,b,c,d},x] && ZeroQ[c-2*a] && ZeroQ[d-2*b]
```

```
Int[Cos[a_.+b_.*x_]/Sqrt[Sin[c_.+d_.*x_]],x_Symbol] :=
  -ArcSin[Cos[a+b*x]-Sin[a+b*x]]/(2*b) + Log[Cos[a+b*x]+Sin[a+b*x]+Sqrt[Sin[c+d*x]]]/(2*b) /;
FreeQ[{a,b,c,d},x] && ZeroQ[c-2*a] && ZeroQ[d-2*b]
```

- **Rule:**

$$\int \frac{\mathrm{Sin}[a+b\,x]}{\mathrm{Sin}[2\,(a+b\,x)]^{3/2}}\,dx \longrightarrow \frac{\mathrm{Sin}[a+b\,x]}{b\,\sqrt{\mathrm{Sin}[2\,(a+b\,x)]}}$$

- **Program code:**

```
Int[Sin[a_.+b_.*x_]/Sin[c_.+d_.*x_]^(3/2),x_Symbol] :=
  Sin[a+b*x]/(b*Sqrt[Sin[c+d*x]]) /;
FreeQ[{a,b,c,d},x] && ZeroQ[c-2*a] && ZeroQ[d-2*b]
```

```
Int[Cos[a_.+b_.*x_]/Sin[c_.+d_.*x_]^(3/2),x_Symbol] :=
  -Cos[a+b*x]/(b*Sqrt[Sin[c+d*x]]) /;
FreeQ[{a,b,c,d},x] && ZeroQ[c-2*a] && ZeroQ[d-2*b]
```

- **Rule: If $n > 0$, then**

$$\int \text{Sin}[a+b\,x]\,\text{Sin}[2\,(a+b\,x)]^n\,dx \longrightarrow$$
$$-\frac{\text{Cos}[a+b\,x]\,\text{Sin}[2\,(a+b\,x)]^n}{b\,(2\,n+1)}+\frac{2\,n}{2\,n+1}\int \text{Cos}[a+b\,x]\,\text{Sin}[2\,(a+b\,x)]^{n-1}\,dx$$

- **Program code:**

```
Int[Sin[a_.+b_.*x_]*Sin[c_.+d_.*x_]^n_.,x_Symbol] :=
  -Cos[a+b*x]*Sin[c+d*x]^n/(b*(2*n+1)) +
  Dist[2*n/(2*n+1),Int[Cos[a+b*x]*Sin[c+d*x]^(n-1),x]] /;
FreeQ[{a,b,c,d},x] && ZeroQ[c-2*a] && ZeroQ[d-2*b] && FractionQ[n] && n>0
```

```
Int[Cos[a_.+b_.*x_]*Sin[c_.+d_.*x_]^n_.,x_Symbol] :=
  Sin[a+b*x]*Sin[c+d*x]^n/(b*(2*n+1)) +
  Dist[2*n/(2*n+1),Int[Sin[a+b*x]*Sin[c+d*x]^(n-1),x]] /;
FreeQ[{a,b,c,d},x] && ZeroQ[c-2*a] && ZeroQ[d-2*b] && FractionQ[n] && n>0
```

- **Rule: If $n < -1 \bigwedge n \neq -\frac{3}{2}$, then**

$$\int \text{Sin}[a+b\,x]\,\text{Sin}[2\,(a+b\,x)]^n\,dx \longrightarrow$$
$$-\frac{\text{Sin}[a+b\,x]\,\text{Sin}[2\,(a+b\,x)]^{n+1}}{2\,b\,(n+1)}+\frac{2\,n+3}{2\,(n+1)}\int \text{Cos}[a+b\,x]\,\text{Sin}[2\,(a+b\,x)]^{n+1}\,dx$$

- **Program code:**

```
Int[Sin[a_.+b_.*x_]*Sin[c_.+d_.*x_]^n_,x_Symbol] :=
  -Sin[a+b*x]*Sin[c+d*x]^(n+1)/(2*b*(n+1)) +
  Dist[(2*n+3)/(2*(n+1)),Int[Cos[a+b*x]*Sin[c+d*x]^(n+1),x]] /;
FreeQ[{a,b,c,d},x] && ZeroQ[c-2*a] && ZeroQ[d-2*b] && FractionQ[n] && n<-1 && n≠-3/2
```

```
Int[Cos[a_.+b_.*x_]*Sin[c_.+d_.*x_]^n_,x_Symbol] :=
  Cos[a+b*x]*Sin[c+d*x]^(n+1)/(2*b*(n+1)) +
  Dist[(2*n+3)/(2*(n+1)),Int[Sin[a+b*x]*Sin[c+d*x]^(n+1),x]] /;
FreeQ[{a,b,c,d},x] && ZeroQ[c-2*a] && ZeroQ[d-2*b] && FractionQ[n] && n<-1 && n≠-3/2
```

$\int u\,\mathrm{Sin}[v]\,\mathrm{Trig}[w]\,dx$

- **Derivation: Algebraic expansion**
- **Basis:** $\mathrm{Sin}[v]\,\mathrm{Cos}[w] = \frac{1}{2}\mathrm{Sin}[v+w] + \frac{1}{2}\mathrm{Sin}[v-w]$
- **Rule: If** $v,\ w \in \mathbb{P}x \land v+w \neq 0 \land v-w \neq 0$**, then**

$$\int u\,\mathrm{Sin}[v]\,\mathrm{Cos}[w]\,dx \longrightarrow \frac{1}{2}\int u\,\mathrm{Sin}[v+w]\,dx + \frac{1}{2}\int u\,\mathrm{Sin}[v-w]\,dx$$

- **Program code:**

```
Int[u_.*Sin[v_]*Cos[w_],x_Symbol] :=
  Dist[1/2,Int[u*Regularize[Sin[v+w],x],x]] +
  Dist[1/2,Int[u*Regularize[Sin[v-w],x],x]] /;
(PolynomialQ[v,x] && PolynomialQ[w,x] || IndependentQ[Cancel[v/w],x]) && NonzeroQ[v+w] && NonzeroQ[v-
```

- **Derivation: Algebraic expansion**
- **Basis:** $\mathrm{Sin}[v]\,\mathrm{Sin}[w] = \frac{1}{2}\mathrm{Cos}[v-w] - \frac{1}{2}\mathrm{Cos}[v+w]$
- **Rule: If** $v,\ w \in \mathbb{P}x \land v+w \neq 0 \land v-w \neq 0$**, then**

$$\int u\,\mathrm{Sin}[v]\,\mathrm{Sin}[w]\,dx \longrightarrow \frac{1}{2}\int u\,\mathrm{Cos}[v-w]\,dx - \frac{1}{2}\int u\,\mathrm{Cos}[v+w]\,dx$$

- **Program code:**

```
Int[u_.*Sin[v_]*Sin[w_],x_Symbol] :=
  Dist[1/2,Int[u*Regularize[Cos[v-w],x],x]] -
  Dist[1/2,Int[u*Regularize[Cos[v+w],x],x]] /;
(PolynomialQ[v,x] && PolynomialQ[w,x] || IndependentQ[Cancel[v/w],x]) && NonzeroQ[v+w] && NonzeroQ[v-
```

- **Basis:** $\mathrm{Cos}[v]\,\mathrm{Cos}[w] = \frac{1}{2}\mathrm{Cos}[v-w] + \frac{1}{2}\mathrm{Cos}[v+w]$

```
Int[u_.*Cos[v_]*Cos[w_],x_Symbol] :=
  Dist[1/2,Int[u*Regularize[Cos[v-w],x],x]] +
  Dist[1/2,Int[u*Regularize[Cos[v+w],x],x]] /;
(PolynomialQ[v,x] && PolynomialQ[w,x] || IndependentQ[Cancel[v/w],x]) && NonzeroQ[v+w] && NonzeroQ[v-
```

- **Derivation: Algebraic expansion**
- **Basis:** $\text{Sin}[v]\ \text{Tan}[w] = -\text{Cos}[v] + \text{Cos}[v-w]\ \text{Sec}[w]$
- **Rule: If** $n > 0 \wedge x \notin v - w \neq 0$**, then**

$$\int u\,\text{Sin}[v]\ \text{Tan}[w]^n\,dx \longrightarrow -\int u\,\text{Cos}[v]\ \text{Tan}[w]^{n-1}\,dx + \text{Cos}[v-w]\int u\,\text{Sec}[w]\ \text{Tan}[w]^{n-1}\,dx$$

- **Program code:**

```
Int[u_.*Sin[v_]*Tan[w_]^n_.,x_Symbol] :=
  -Int[u*Cos[v]*Tan[w]^(n-1),x] + Cos[v-w]*Int[u*Sec[w]*Tan[w]^(n-1),x] /;
RationalQ[n] && n>0 && FreeQ[v-w,x] && NonzeroQ[v-w]
```

- **Basis:** $\text{Cos}[v]\ \text{Cot}[w] = -\text{Sin}[v] + \text{Cos}[v-w]\ \text{Csc}[w]$

```
Int[u_.*Cos[v_]*Cot[w_]^n_.,x_Symbol] :=
  -Int[u*Sin[v]*Cot[w]^(n-1),x] + Cos[v-w]*Int[u*Csc[w]*Cot[w]^(n-1),x] /;
RationalQ[n] && n>0 && FreeQ[v-w,x] && NonzeroQ[v-w]
```

- **Derivation: Algebraic expansion**
- **Basis:** $\text{Sin}[v]\ \text{Cot}[w] = \text{Cos}[v] + \text{Sin}[v-w]\ \text{Csc}[w]$
- **Rule: If** $n > 0 \wedge x \notin v - w \neq 0$**, then**

$$\int u\,\text{Sin}[v]\ \text{Cot}[w]^n\,dx \longrightarrow \int u\,\text{Cos}[v]\ \text{Cot}[w]^{n-1}\,dx + \text{Sin}[v-w]\int u\,\text{Csc}[w]\ \text{Cot}[w]^{n-1}\,dx$$

- **Program code:**

```
Int[u_.*Sin[v_]*Cot[w_]^n_.,x_Symbol] :=
  Int[u*Cos[v]*Cot[w]^(n-1),x] + Sin[v-w]*Int[u*Csc[w]*Cot[w]^(n-1),x] /;
RationalQ[n] && n>0 && FreeQ[v-w,x] && NonzeroQ[v-w]
```

- **Basis:** $\text{Cos}[v]\ \text{Tan}[w] = \text{Sin}[v] - \text{Sin}[v-w]\ \text{Sec}[w]$

```
Int[u_.*Cos[v_]*Tan[w_]^n_.,x_Symbol] :=
  Int[u*Sin[v]*Tan[w]^(n-1),x] - Sin[v-w]*Int[u*Sec[w]*Tan[w]^(n-1),x] /;
RationalQ[n] && n>0 && FreeQ[v-w,x] && NonzeroQ[v-w]
```

- **Derivation: Algebraic expansion**
- **Basis:** $\text{Sin}[v]\ \text{Sec}[w] = \text{Cos}[v-w]\ \text{Tan}[w] + \text{Sin}[v-w]$
- **Rule: If** $n>0 \land x \notin v-w \neq 0$**, then**

$$\int u\ \text{Sin}[v]\ \text{Sec}[w]^n\, dx \longrightarrow \text{Cos}[v-w] \int u\ \text{Tan}[w]\ \text{Sec}[w]^{n-1}\, dx + \text{Sin}[v-w] \int u\ \text{Sec}[w]^{n-1}\, dx$$

- **Program code:**

```
Int[u_.*Sin[v_]*Sec[w_]^n_.,x_Symbol] :=
  Cos[v-w]*Int[u*Tan[w]*Sec[w]^(n-1),x] + Sin[v-w]*Int[u*Sec[w]^(n-1),x] /;
RationalQ[n] && n>0 && FreeQ[v-w,x] && NonzeroQ[v-w]
```

- **Basis:** $\text{Cos}[v] * \text{Csc}[w] = \text{Cos}[v-w] * \text{Cot}[w] - \text{Sin}[v-w]$

```
Int[u_.*Cos[v_]*Csc[w_]^n_.,x_Symbol] :=
  Cos[v-w]*Int[u*Cot[w]*Csc[w]^(n-1),x] - Sin[v-w]*Int[u*Csc[w]^(n-1),x] /;
RationalQ[n] && n>0 && FreeQ[v-w,x] && NonzeroQ[v-w]
```

- **Derivation: Algebraic expansion**
- **Basis:** $\text{Sin}[v]\ \text{Csc}[w] = \text{Sin}[v-w]\ \text{Cot}[w] + \text{Cos}[v-w]$
- **Rule: If** $n>0 \land x \notin v-w \neq 0$**, then**

$$\int u\ \text{Sin}[v]\ \text{Csc}[w]^n\, dx \longrightarrow \text{Sin}[v-w] \int u\ \text{Cot}[w]\ \text{Csc}[w]^{n-1}\, dx + \text{Cos}[v-w] \int u\ \text{Csc}[w]^{n-1}\, dx$$

- **Program code:**

```
Int[u_.*Sin[v_]*Csc[w_]^n_.,x_Symbol] :=
  Sin[v-w]*Int[u*Cot[w]*Csc[w]^(n-1),x] + Cos[v-w]*Int[u*Csc[w]^(n-1),x] /;
RationalQ[n] && n>0 && FreeQ[v-w,x] && NonzeroQ[v-w]
```

- **Basis:** $\text{Cos}[v]\ \text{Sec}[w] = -\text{Sin}[v-w]\ \text{Tan}[w] + \text{Cos}[v-w]$

```
Int[u_.*Cos[v_]*Sec[w_]^n_.,x_Symbol] :=
  -Sin[v-w]*Int[u*Tan[w]*Sec[w]^(n-1),x] + Cos[v-w]*Int[u*Sec[w]^(n-1),x] /;
RationalQ[n] && n>0 && FreeQ[v-w,x] && NonzeroQ[v-w]
```

$$\int x^m \operatorname{Sin}[a + b\, x^n]^p \operatorname{Cos}[a + b\, x^n]\, dx$$

- **Reference: G&R 2.645.6**
- **Rule: If $m, n, p \in \mathbb{Z} \wedge p \neq -1 \wedge 0 < n \leq m$, then**

$$\int x^m \operatorname{Sin}[a + b\, x^n]^p \operatorname{Cos}[a + b\, x^n]\, dx \longrightarrow \frac{x^{m-n+1} \operatorname{Sin}[a + b\, x^n]^{p+1}}{b\, n\, (p+1)} - \frac{m-n+1}{b\, n\, (p+1)} \int x^{m-n} \operatorname{Sin}[a + b\, x^n]^{p+1}\, dx$$

- **Program code:**

```
Int[x_^m_.*Sin[a_.+b_.*x_^n_.]^p_.*Cos[a_.+b_.*x_^n_.],x_Symbol] :=
  x^(m-n+1)*Sin[a+b*x^n]^(p+1)/(b*n*(p+1)) -
  Dist[(m-n+1)/(b*n*(p+1)),Int[x^(m-n)*Sin[a+b*x^n]^(p+1),x]] /;
FreeQ[{a,b},x] && IntegersQ[m,n,p] && p≠-1 && 0<n≤m
```

- **Reference: G&R 2.645.3**

```
Int[x_^m_.*Cos[a_.+b_.*x_^n_.]^p_.*Sin[a_.+b_.*x_^n_.],x_Symbol] :=
  -x^(m-n+1)*Cos[a+b*x^n]^(p+1)/(b*n*(p+1)) +
  Dist[(m-n+1)/(b*n*(p+1)),Int[x^(m-n)*Cos[a+b*x^n]^(p+1),x]] /;
FreeQ[{a,b},x] && IntegersQ[m,n,p] && p≠-1 && 0<n≤m
```

$$\int \mathrm{Sin}[a+b\,x]^m\,\mathrm{Cos}[a+b\,x]^n\,dx$$

- **Rule: If** $m+n+2=0\,\wedge\,m+1\neq 0$**, then**

$$\int \mathrm{Sin}[a+b\,x]^m\,\mathrm{Cos}[a+b\,x]^n\,dx\;\longrightarrow\;\frac{\mathrm{Sin}[a+b\,x]^{m+1}\,\mathrm{Cos}[a+b\,x]^{n+1}}{b\,(m+1)}$$

- **Program code:**

```
Int[Sin[a_.+b_.*x_]^m_.*Cos[a_.+b_.*x_]^n_,x_Symbol] :=
  Sin[a+b*x]^(m+1)*Cos[a+b*x]^(n+1)/(b*(m+1)) /;
FreeQ[{a,b,m,n},x] && ZeroQ[m+n+2] && NonzeroQ[m+1]
```

- **Derivation: Integration by substitution**
- **Basis: If** $\frac{n-1}{2}\in\mathbb{Z}$**, then** $\mathrm{Sin}[z]^m\,\mathrm{Cos}[z]^n=\mathrm{Sin}[z]^m\left(1-\mathrm{Sin}[z]^2\right)^{\frac{n-1}{2}}\mathrm{Sin}'[z]$
- **Note: This rule is used for odd** n **since it requires fewer steps and results in a simpler antiderivative than the other rules in this section.**
- **Rule: If** $\frac{n-1}{2}\in\mathbb{Z}\;\bigwedge\;\neg\left(\frac{m-1}{2}\in\mathbb{Z}\;\bigwedge\;0<m<n\right)$**, then**

$$\int \mathrm{Sin}[a+b\,x]^m\,\mathrm{Cos}[a+b\,x]^n\,dx\;\longrightarrow\;\frac{1}{b}\,\mathrm{Subst}\left[\int x^m\left(1-x^2\right)^{\frac{n-1}{2}}dx,\,x,\,\mathrm{Sin}[a+b\,x]\right]$$

- **Program code:**

```
Int[Sin[a_.+b_.*x_]^m_*Cos[a_.+b_.*x_]^n_,x_Symbol] :=
  Dist[1/b,Subst[Int[Regularize[x^m*(1-x^2)^((n-1)/2),x],x],x,Sin[a+b*x]]] /;
FreeQ[{a,b,m},x] && OddQ[n] && Not[OddQ[m] && 0<m<n]
```

- **Basis: If** $\frac{m-1}{2}\in\mathbb{Z}$**, then** $\mathrm{Sin}[z]^m\,\mathrm{Cos}[z]^n=-\mathrm{Cos}[z]^n\left(1-\mathrm{Cos}[z]^2\right)^{\frac{m-1}{2}}\mathrm{Cos}'[z]$

```
Int[Sin[a_.+b_.*x_]^m_*Cos[a_.+b_.*x_]^n_,x_Symbol] :=
  Dist[-1/b,Subst[Int[Regularize[x^n*(1-x^2)^((m-1)/2),x],x],x,Cos[a+b*x]]] /;
FreeQ[{a,b,n},x] && OddQ[m] && Not[OddQ[n] && 0<n<=m]
```

- **Reference: G&R 2.510.1**
- **Rule: If** $m > 1 \land n < -1$, **then**

$$\int \mathrm{Sin}[a+b\,x]^m\,\mathrm{Cos}[a+b\,x]^n\,dx \longrightarrow -\frac{\mathrm{Sin}[a+b\,x]^{m-1}\,\mathrm{Cos}[a+b\,x]^{n+1}}{b\,(n+1)}+\frac{m-1}{n+1}\int \mathrm{Sin}[a+b\,x]^{m-2}\,\mathrm{Cos}[a+b\,x]^{n+2}\,dx$$

- **Program code:**

```
Int[Sin[a_.+b_.*x_]^m_*Cos[a_.+b_.*x_]^n_,x_Symbol] :=
  -Sin[a+b*x]^(m-1)*Cos[a+b*x]^(n+1)/(b*(n+1)) +
  Dist[(m-1)/(n+1),Int[Sin[a+b*x]^(m-2)*Cos[a+b*x]^(n+2),x]] /;
FreeQ[{a,b},x] && RationalQ[{m,n}] && m>1 && n<-1
```

- **Reference: G&R 2.510.4**

```
Int[Sin[a_.+b_.*x_]^m_*Cos[a_.+b_.*x_]^n_,x_Symbol] :=
  Sin[a+b*x]^(m+1)*Cos[a+b*x]^(n-1)/(b*(m+1)) +
  Dist[(n-1)/(m+1),Int[Sin[a+b*x]^(m+2)*Cos[a+b*x]^(n-2),x]] /;
FreeQ[{a,b},x] && RationalQ[{m,n}] && m<-1 && n>1
```

- **Reference: G&R 2.510.2, CRC 323b, A&S 4.3.127b**
- **Rule: If** $m > 1 \bigwedge \frac{m-1}{2} \notin \mathbb{Z} \bigwedge m+n \neq 0 \bigwedge \neg\left(\frac{n-1}{2} \in \mathbb{Z} \bigwedge n > 1\right)$, **then**

$$\int \mathrm{Sin}[a+b\,x]^m\,\mathrm{Cos}[a+b\,x]^n\,dx \longrightarrow -\frac{\mathrm{Sin}[a+b\,x]^{m-1}\,\mathrm{Cos}[a+b\,x]^{n+1}}{b\,(m+n)}+\frac{m-1}{m+n}\int \mathrm{Sin}[a+b\,x]^{m-2}\,\mathrm{Cos}[a+b\,x]^{n}\,dx$$

- **Program code:**

```
Int[Sin[a_.+b_.*x_]^m_*Cos[a_.+b_.*x_]^n_,x_Symbol] :=
  -Sin[a+b*x]^(m-1)*Cos[a+b*x]^(n+1)/(b*(m+n)) +
  Dist[(m-1)/(m+n),Int[Sin[a+b*x]^(m-2)*Cos[a+b*x]^n,x]] /;
FreeQ[{a,b,n},x] && RationalQ[m] && m>1 && NonzeroQ[m+n]
```

- **Reference: G&R 2.510.5, CRC 323a, A&S 4.3.127a**

```
Int[Sin[a_.+b_.*x_]^m_*Cos[a_.+b_.*x_]^n_,x_Symbol] :=
  Sin[a+b*x]^(m+1)*Cos[a+b*x]^(n-1)/(b*(m+n)) +
  Dist[(n-1)/(m+n),Int[Sin[a+b*x]^m*Cos[a+b*x]^(n-2),x]] /;
FreeQ[{a,b,m},x] && RationalQ[n] && n>1 && NonzeroQ[m+n]
```

- **Reference: G&R 2.510.3, CRC 334a, A&S 4.3.128b**
- **Rule: If $m < -1 \land m + n + 2 \neq 0$, then**

$$\int \mathrm{Sin}[a+b\,x]^m\,\mathrm{Cos}[a+b\,x]^n\,dx \longrightarrow \frac{\mathrm{Sin}[a+b\,x]^{m+1}\,\mathrm{Cos}[a+b\,x]^{n+1}}{b\,(m+1)} + \frac{m+n+2}{m+1}\int \mathrm{Sin}[a+b\,x]^{m+2}\,\mathrm{Cos}[a+b\,x]^n\,dx$$

- **Program code:**

```
Int[Sin[a_.+b_.*x_]^m_*Cos[a_.+b_.*x_]^n_,x_Symbol] :=
  Sin[a+b*x]^(m+1)*Cos[a+b*x]^(n+1)/(b*(m+1)) +
  Dist[(m+n+2)/(m+1),Int[Sin[a+b*x]^(m+2)*Cos[a+b*x]^n,x]] /;
FreeQ[{a,b,n},x] && RationalQ[m] && m<-1 && NonzeroQ[m+n+2]
```

- **Reference: G&R 2.510.6, CRC 334b, A&S 4.3.128a**

```
Int[Sin[a_.+b_.*x_]^m_*Cos[a_.+b_.*x_]^n_,x_Symbol] :=
  -Sin[a+b*x]^(m+1)*Cos[a+b*x]^(n+1)/(b*(n+1)) +
  Dist[(m+n+2)/(n+1),Int[Sin[a+b*x]^m*Cos[a+b*x]^(n+2),x]] /;
FreeQ[{a,b,m},x] && RationalQ[n] && n<-1 && NonzeroQ[m+n+2]
```

- **Derivation: Integration by substitution**
- **Basis: If $\frac{1}{m} \in \mathbb{Z}$, then** $\frac{\mathrm{Sin}[z]^m}{\mathrm{Cos}[z]^m} = \frac{\left(\frac{\mathrm{Sin}[z]^m}{\mathrm{Cos}[z]^m}\right)^{1/m}}{m\left(1+\left(\frac{\mathrm{Sin}[z]^m}{\mathrm{Cos}[z]^m}\right)^{2/m}\right)}\,\partial_z \frac{\mathrm{Sin}[z]^m}{\mathrm{Cos}[z]^m}$
- **Note: This rule should be replaced with a more general one.**
- **Rule: If $\frac{1}{m} \in \mathbb{Z} \bigwedge \frac{1}{m} > 1$, then**

$$\int \frac{\mathrm{Sin}[a+b\,x]^m}{\mathrm{Cos}[a+b\,x]^m}\,dx \longrightarrow \frac{1}{b\,m}\,\mathrm{Subst}\left[\int \frac{x^{1/m}}{1+x^{2/m}}\,dx,\ x,\ \frac{\mathrm{Sin}[a+b\,x]^m}{\mathrm{Cos}[a+b\,x]^m}\right]$$

- **Program code:**

```
Int[Sin[a_.+b_.*x_]^m_*Cos[a_.+b_.*x_]^n_,x_Symbol] :=
  Dist[1/(b*m),Subst[Int[x^(1/m)/(1+x^(2/m)),x],x,Sin[a+b*x]^m/Cos[a+b*x]^m]] /;
FreeQ[{a,b},x] && ZeroQ[m+n] && IntegerQ[1/m] && 1/m>1
```

- **Basis: If $\frac{1}{n} \in \mathbb{Z}$, then** $\frac{\mathrm{Cos}[z]^n}{\mathrm{Sin}[z]^n} = -\frac{\left(\frac{\mathrm{Cos}[z]^n}{\mathrm{Sin}[z]^n}\right)^{1/n}}{n\left(1+\left(\frac{\mathrm{Cos}[z]^n}{\mathrm{Sin}[z]^n}\right)^{2/n}\right)}\,\partial_z \frac{\mathrm{Cos}[z]^n}{\mathrm{Sin}[z]^n}$

```
Int[Sin[a_.+b_.*x_]^m_*Cos[a_.+b_.*x_]^n_,x_Symbol] :=
  Dist[-1/(b*n),Subst[Int[x^(1/n)/(1+x^(2/n)),x],x,Cos[a+b*x]^n/Sin[a+b*x]^n]] /;
FreeQ[{a,b},x] && ZeroQ[m+n] && IntegerQ[1/n] && 1/n>1
```

$$\int x^m \left(a + b \operatorname{Cos}[d + e x]^2 + c \operatorname{Sin}[d + e x]^2\right)^n dx$$

- **Derivation: Algebraic simplification**
- **Basis:** $a + b \operatorname{Cos}[z]^2 + c \operatorname{Sin}[z]^2 = \frac{1}{2} (2 a + b + c + (b - c) \operatorname{Cos}[2 z])$
- **Rule: If** $m \in \mathbb{Z} \land m > 0 \land a + b \neq 0 \land a + c \neq 0$**, then**

$$\int \frac{x^m}{a + b \operatorname{Cos}[d + e x]^2 + c \operatorname{Sin}[d + e x]^2} dx \longrightarrow 2 \int \frac{x^m}{2 a + b + c + (b - c) \operatorname{Cos}[2 d + 2 e x]} dx$$

- **Program code:**

```
Int[x_^m_./(a_.+b_.*Cos[d_.+e_.*x_]^2+c_.*Sin[d_.+e_.*x_]^2),x_Symbol] :=
  Dist[2,Int[x^m/(2*a+b+c+(b-c)*Cos[2*d+2*e*x]),x]] /;
FreeQ[{a,b,c,d,e},x] && IntegerQ[m] && m>0 && NonzeroQ[a+b] && NonzeroQ[a+c]
```

$$\int x^m\,(a+b\,\mathrm{Sin}[c+d\,x]\,\mathrm{Cos}[c+d\,x])^n\,dx$$

- **Derivation: Algebraic simplification**
- **Basis:** $\mathrm{Sin}[z]\,\mathrm{Cos}[z]=\frac{1}{2}\,\mathrm{Sin}[2\,z]$
- **Rule: If** $m\in\mathbb{Z}\land m>0$**, then**

$$\int\frac{x^m}{a+b\,\mathrm{Sin}[c+d\,x]\,\mathrm{Cos}[c+d\,x]}\,dx\longrightarrow\int\frac{x^m}{a+\frac{1}{2}\,b\,\mathrm{Sin}[2\,c+2\,d\,x]}\,dx$$

- **Program code:**

```
Int[x_^m_./(a_+b_.*Sin[c_.+d_.*x_]*Cos[c_.+d_.*x_]),x_Symbol] :=
  Int[x^m/(a+b*Sin[2*c+2*d*x]/2),x] /;
FreeQ[{a,b,c,d},x] && IntegerQ[m] && m>0
```

- **Derivation: Algebraic simplification**
- **Basis:** $\mathrm{Sin}[z]\,\mathrm{Cos}[z]=\frac{1}{2}\,\mathrm{Sin}[2\,z]$
- **Rule: If** $n-\frac{1}{2}\in\mathbb{Z}$**, then**

$$\int(a+b\,\mathrm{Sin}[c+d\,x]\,\mathrm{Cos}[c+d\,x])^n\,dx\longrightarrow\int\left(a+\frac{1}{2}\,b\,\mathrm{Sin}[2\,c+2\,d\,x]\right)^n\,dx$$

- **Program code:**

```
Int[(a_+b_.*Sin[c_.+d_.*x_]*Cos[c_.+d_.*x_])^n_,x_Symbol] :=
  Int[(a+b*Sin[2*c+2*d*x]/2)^n,x] /;
FreeQ[{a,b,c,d},x] && HalfIntegerQ[n]
```

$$\int \text{Sin}[a + b\, x^n]^p\, \text{Cos}[a + b\, x^n]^p\, dx$$

- **Derivation: Algebraic simplification**
- **Basis:** $\text{Sin}[z]\, \text{Cos}[z] = \frac{1}{2} \text{Sin}[2\, z]$
- **Rule: If** $n,\ p \in \mathbb{Z}$**, then**

$$\int \text{Sin}[a + b\, x^n]^p\, \text{Cos}[a + b\, x^n]^p\, dx \longrightarrow \frac{1}{2} \int \text{Sin}[2\, a + 2\, b\, x^n]^p\, dx$$

- **Program code:**

```
Int[Sin[a_.+b_.*x_^n_]^p_.*Cos[a_.+b_.*x_^n_]^p_.,x_Symbol] :=
  Dist[1/2,Int[Sin[2*a+2*b*x^n]^p,x]] /;
FreeQ[{a,b},x] && IntegersQ[n,p]
```

$$\int (a \operatorname{Csc}[c + d x] + b \operatorname{Sin}[c + d x])^n \, dx$$

- **Derivation: Algebraic simplification**
- **Basis:** $\operatorname{Csc}[z] - \operatorname{Sin}[z] = \operatorname{Cos}[z] \operatorname{Cot}[z]$
- **Rule: If** $a + b = 0$, **then**

$$\int (a \operatorname{Csc}[c + d x] + b \operatorname{Sin}[c + d x])^n \, dx \longrightarrow \int (a \operatorname{Cos}[c + d x] \operatorname{Cot}[c + d x])^n \, dx$$

- **Program code:**

```
Int[(a_.*Csc[c_.+d_.*x_]+b_.*Sin[c_.+d_.*x_])^n_,x_Symbol] :=
  Int[(a*Cos[c+d*x]*Cot[c+d*x])^n,x] /;
FreeQ[{a,b,c,d,n},x] && ZeroQ[a+b]
```

```
Int[(a_.*Sec[c_.+d_.*x_]+b_.*Cos[c_.+d_.*x_])^n_,x_Symbol] :=
  Int[(a*Sin[c+d*x]*Tan[c+d*x])^n,x] /;
FreeQ[{a,b,c,d,n},x] && ZeroQ[a+b]
```

$$\int \text{Sec}[v]^m\,(a+b\,\text{Tan}[v])^n\,dx$$

- **Derivation: Algebraic simplification**
- **Basis:** $\frac{a+b\,\text{Tan}[z]}{\text{Sec}[z]} = a\,\text{Cos}[z] + b\,\text{Sin}[z]$
- **Rule: If** $m,\ n \in \mathbb{Z}\ \land\ m+n=0\ \bigwedge\ \frac{m-1}{2} \in \mathbb{Z}$**, then**

$$\int \text{Sec}[v]^m\,(a+b\,\text{Tan}[v])^n\,dx \longrightarrow \int (a\,\text{Cos}[v]+b\,\text{Sin}[v])^n\,dx$$

- **Program code:**

```
Int[Sec[v_]^m_.*(a_+b_.*Tan[v_])^n_., x_Symbol] :=
  Int[(a*Cos[v]+b*Sin[v])^n,x] /;
FreeQ[{a,b},x] && IntegersQ[m,n] && m+n==0 && OddQ[m]
```

```
Int[Csc[v_]^m_.*(a_+b_.*Cot[v_])^n_., x_Symbol] :=
  Int[(b*Cos[v]+a*Sin[v])^n,x] /;
FreeQ[{a,b},x] && IntegersQ[m,n] && m+n==0 && OddQ[m]
```

$$\int x^m \operatorname{Csc}[a+b\,x]^n \operatorname{Sec}[a+b\,x]^p \,dx$$

- **Derivation: Algebraic simplification**
- **Basis:** $\operatorname{Csc}[z]\operatorname{Sec}[z] = 2\operatorname{Csc}[2\,z]$
- **Rule: If** $n \in \mathbb{Z}$**, then**

$$\int x^m \operatorname{Csc}[a+b\,x]^n \operatorname{Sec}[a+b\,x]^n \,dx \longrightarrow 2^n \int x^m \operatorname{Csc}[2\,a+2\,b\,x]^n \,dx$$

- **Program code:**

```
Int[x_^m_.*Csc[a_.+b_.*x_]^n_.*Sec[a_.+b_.*x_]^n_., x_Symbol] :=
  Dist[2^n,Int[x^m*Csc[2*a+2*b*x]^n,x]] /;
FreeQ[{a,b},x] && RationalQ[m] && IntegerQ[n]
```

- **Derivation: Integration by parts**
- **Rule: If** $n,\ p \in \mathbb{Z} \land m > 0 \land n \neq p$**, then**

$$\int x^m \operatorname{Csc}[a+b\,x]^n \operatorname{Sec}[a+b\,x]^p \,dx \longrightarrow$$
$$x^m \int \operatorname{Csc}[a+b\,x]^n \operatorname{Sec}[a+b\,x]^p \,dx - m \int x^{m-1} \left(\int \operatorname{Csc}[a+b\,x]^n \operatorname{Sec}[a+b\,x]^p \,dx\right) dx$$

- **Program code:**

```
Int[x_^m_.*Csc[a_.+b_.*x_]^n_.*Sec[a_.+b_.*x_]^p_., x_Symbol] :=
  Module[{u=Block[{ShowSteps=False,StepCounter=Null}, Int[Csc[a+b*x]^n*Sec[a+b*x]^p,x]]},
  x^m*u - Dist[m,Int[x^(m-1)*u,x]]] /;
FreeQ[{a,b},x] && RationalQ[m] && IntegersQ[n,p] && m>0 && n≠p
```

$$\int u\,(a\,\mathrm{Tan}[v]^m+b\,\mathrm{Sec}[v]^m)^n\,dx$$

- **Derivation: Algebraic simplification**
- **Basis: If** $a^2-b^2=0$, **then** $a\,\mathrm{Tan}[z]+b\,\mathrm{Sec}[z]=a\,\mathrm{Tan}\left[\frac{z}{2}+\frac{a\,\pi}{b\,4}\right]$
- **Rule: If** $a^2-b^2=0\ \bigwedge\ \frac{n}{2}\in\mathbb{Z}$, **then**

$$\int(a\,\mathrm{Tan}[v]+b\,\mathrm{Sec}[v])^n\,dx\ \longrightarrow\ a^n\int\mathrm{Tan}\left[\frac{v}{2}+\frac{\pi}{4}\,\frac{a}{b}\right]^n\,dx$$

- **Program code:**

```
Int[(a_.*Tan[v_]+b_.*Sec[v_])^n_,x_Symbol] :=
  Dist[a^n,Int[Tan[v/2+Pi/4*a/b]^n,x]] /;
FreeQ[{a,b},x] && ZeroQ[a^2-b^2] && EvenQ[n]
```

```
Int[(a_.*Cot[v_]+b_.*Csc[v_])^n_,x_Symbol] :=
  Dist[a^n,Int[Cot[v/2+(a/b-1)*Pi/4]^n,x]] /;
FreeQ[{a,b},x] && ZeroQ[a^2-b^2] && EvenQ[n]
```

- **Derivation: Algebraic simplification**
- **Basis:** $a\,\mathrm{Sec}[z]+b\,\mathrm{Tan}[z]=\frac{a+b\,\mathrm{Sin}[z]}{\mathrm{Cos}[z]}$
- **Rule: If** $m,\ n\in\mathbb{Z}\ \bigwedge\ \left(\frac{m\,n-1}{2}\in\mathbb{Z}\ \bigvee\ m\,n<0\right)\ \bigwedge\ \neg\,(m=2\ \wedge\ a+b=0)$, **then**

$$\int u\,(a\,\mathrm{Tan}[v]^m+b\,\mathrm{Sec}[v]^m)^n\,dx\ \longrightarrow\ \int\frac{u\,(a+b\,\mathrm{Sin}[v]^m)^n}{\mathrm{Cos}[v]^{m\,n}}\,dx$$

- **Program code:**

```
Int[u_.*(a_.*Sec[v_]^m_.+b_.*Tan[v_]^m_.)^n_.,x_Symbol] :=
  Int[u*(a+b*Sin[v]^m)^n/Cos[v]^(m*n),x] /;
FreeQ[{a,b},x] && IntegersQ[m,n] && (OddQ[m*n] || m*n<0) && Not[m==2 && ZeroQ[a+b]]
```

```
Int[u_.*(a_.*Csc[v_]^m_.+b_.*Cot[v_]^m_.)^n_.,x_Symbol] :=
  Int[u*(a+b*Cos[v]^m)^n/Sin[v]^(m*n),x] /;
FreeQ[{a,b},x] && IntegersQ[m,n] && (OddQ[m*n] || m*n<0) && Not[m==2 && ZeroQ[a+b]]
```