$\int f^{a+b\,x^n}\,dx$

- **Reference: G&R 2.311, CRC 519, A&S 4.2.54**
- **Derivation: Primitive rule**
- **Basis:** $\partial_x e^x = e^x$
- **Rule:**

$$\int f^{a+b\,x}\,dx \longrightarrow \frac{f^{a+b\,x}}{b\,\mathrm{Log}[f]}$$

- **Program code:**

```
Int[E^(a_.+b_.*x_),x_Symbol] :=
  E^(a+b*x)/b /;
FreeQ[{a,b},x]
```

```
Int[f_^(a_.+b_.*x_),x_Symbol] :=
  f^(a+b*x)/(b*Log[f]) /;
FreeQ[{a,b,f},x]
```

- **Derivation: Primitive rule**
- **Basis:** $\mathrm{Erfi}'[z] = \frac{2\,e^{z^2}}{\sqrt{\pi}}$
- **Rule: If** $b > 0$, **then**

$$\int f^{a+b\,x^2}\,dx \longrightarrow \frac{f^a\,\sqrt{\pi}\,\mathrm{Erfi}\left[x\,\sqrt{b\,\mathrm{Log}[f]}\right]}{2\,\sqrt{b\,\mathrm{Log}[f]}}$$

- **Program code:**

```
Int[E^(a_.+b_.*x_^2),x_Symbol] :=
  E^a*Sqrt[Pi]*Erfi[x*Rt[b,2]]/(2*Rt[b,2]) /;
FreeQ[{a,b},x] && PosQ[b]
```

```
Int[f_^(a_.+b_.*x_^2),x_Symbol] :=
  f^a*Sqrt[Pi]*Erfi[x*Rt[b*Log[f],2]]/(2*Rt[b*Log[f],2]) /;
FreeQ[{a,b,f},x] && PosQ[b*Log[f]]
```

- **Derivation: Primitive rule**
- **Basis:** $\mathrm{Erf}'[z] = \frac{2\, e^{-z^2}}{\sqrt{\pi}}$
- **Rule: If** $\neg\,(b > 0)$, **then**

$$\int f^{a+b\,x^2}\,dx \longrightarrow \frac{f^a \sqrt{\pi}\ \mathrm{Erf}\left[x\sqrt{-b\,\mathrm{Log}[f]}\,\right]}{2\sqrt{-b\,\mathrm{Log}[f]}}$$

- **Program code:**

```
Int[E^(a_.+b_.*x_^2),x_Symbol] :=
  E^a*Sqrt[Pi]*Erf[x*Rt[-b,2]]/(2*Rt[-b,2]) /;
FreeQ[{a,b},x] && NegQ[b]
```

```
Int[f_^(a_.+b_.*x_^2),x_Symbol] :=
  f^a*Sqrt[Pi]*Erf[x*Rt[-b*Log[f],2]]/(2*Rt[-b*Log[f],2]) /;
FreeQ[{a,b,f},x] && NegQ[b*Log[f]]
```

- **Derivation: Primitive rule**
- **Basis:** $\partial_x \Gamma(n, x) = -e^{-x}\, x^{n-1}$
- **Rule: If** $\neg\,(b > 0)$, **then**

$$\int f^{a+b\,x^n}\,dx \longrightarrow -\frac{f^a\, x\, \Gamma\left[\frac{1}{n}, -b\,x^n\,\mathrm{Log}[f]\right]}{n\,(-b\,x^n\,\mathrm{Log}[f])^{1/n}}$$

- **Program code:**

```
Int[E^(a_.+b_.*x_^n_),x_Symbol] :=
  -E^a*x*Gamma[1/n,-b*x^n]/(n*(-b*x^n)^(1/n)) /;
FreeQ[{a,b,n},x] && Not[FractionOrNegativeQ[n]]
```

```
Int[f_^(a_.+b_.*x_^n_),x_Symbol] :=
  -f^a*x*Gamma[1/n,-b*x^n*Log[f]]/(n*(-b*x^n*Log[f])^(1/n)) /;
FreeQ[{a,b,f,n},x] && Not[FractionOrNegativeQ[n]]
```

- **Derivation: Integration by parts**
- **Note: Although resulting integrand looks more complicated than the original, rules for improper binomials rectify it.**
- **Rule: If $n \in \mathbb{Z} \land n < 0$, then**

$$\int f^{a+b x^n} \, dx \longrightarrow x f^{a+b x^n} - b n \operatorname{Log}[f] \int x^n f^{a+b x^n} \, dx$$

- **Program code:**

```
Int[E^(a_.+b_.*x_^n_.),x_Symbol] :=
  x*E^(a+b*x^n) -
  Dist[b*n,Int[x^n*E^(a+b*x^n),x]] /;
FreeQ[{a,b},x] && IntegerQ[n] && n<0
```

```
Int[f_^(a_.+b_.*x_^n_.),x_Symbol] :=
  x*f^(a+b*x^n) -
  Dist[b*n*Log[f],Int[x^n*f^(a+b*x^n),x]] /;
FreeQ[{a,b,f},x] && IntegerQ[n] && n<0
```

$$\int x^m f^{a+b x^n} \, dx$$

- **Derivation: Primitive rule**
- **Basis:** $\text{ExpIntegralEi}'[z] = \frac{e^z}{z}$
- **Rule:**

$$\int \frac{f^{a+b x^n}}{x} \, dx \longrightarrow \frac{f^a \, \text{ExpIntegralEi}[b \, x^n \, \text{Log}[f]]}{n}$$

- **Program code:**

```
Int[f_^(a_.+b_.*x_^n_.)/x_,x_Symbol] :=
  f^a*ExpIntegralEi[b*x^n*Log[f]]/n /;
FreeQ[{a,b,f,n},x]
```

- **Reference: G&R 2.321.1, CRC 521, A&S 4.2.55**
- **Derivation: Integration by parts**
- **Basis:** $x^m f^{a+b x^n} = x^{m-n+1} \left(f^{a+b x^n} x^{n-1}\right)$
- **Rule: If** $n \in \mathbb{Z} \wedge 0 < n \le m$**, then**

$$\int x^m f^{a+b x^n} \, dx \longrightarrow \frac{x^{m-n+1} f^{a+b x^n}}{b \, n \, \text{Log}[f]} - \frac{m-n+1}{b \, n \, \text{Log}[f]} \int x^{m-n} f^{a+b x^n} \, dx$$

- **Program code:**

```
Int[x_^m_.*f_^(a_.+b_.*x_^n_.),x_Symbol] :=
  x^(m-n+1)*f^(a+b*x^n)/(b*n*Log[f]) -
  Dist[(m-n+1)/(b*n*Log[f]),Int[x^(m-n)*f^(a+b*x^n),x]] /;
FreeQ[{a,b,f},x] && IntegerQ[n] && RationalQ[m] && 0<n<=m
```

- **Reference: G&R 2.324.1, CRC 523, A&S 4.2.56**
- **Derivation: Integration by parts**
- **Rule: If** $n \in \mathbb{Z} \land (n > 0 \land m < -1 \lor 0 < -n \leq m + 1$**, then**

$$\int x^m f^{a+b x^n} dx \longrightarrow \frac{x^{m+1} f^{a+b x^n}}{m+1} - \frac{b\, n\, \mathrm{Log}[f]}{m+1} \int x^{m+n} f^{a+b x^n} dx$$

- **Program code:**

```
Int[x_^m_.*f_^(a_.+b_.*x_^n_.),x_Symbol] :=
  x^(m+1)*f^(a+b*x^n)/(m+1) -
  Dist[b*n*Log[f]/(m+1),Int[x^(m+n)*f^(a+b*x^n),x]] /;
FreeQ[{a,b,f},x] && IntegerQ[n] && RationalQ[m] && (n>0 && m<-1 || 0<-n<=m+1)
```

- **Rule: If** $m + 1 \neq 0 \bigwedge m - n + 1 \neq 0 \bigwedge \neg \left(m = -\frac{1}{2} \bigwedge n = 1\right)$**, then**

$$\int x^m f^{a+b x^n} dx \longrightarrow -\frac{f^a x^{m+1} \Gamma\left[\frac{m+1}{n}, -b x^n \mathrm{Log}[f]\right]}{n \left(-b x^n \mathrm{Log}[f]\right)^{\frac{m+1}{n}}}$$

- **Program code:**

```
Int[x_^m_.*f_^(a_.+b_.*x_^n_.),x_Symbol] :=
  -f^a*x^(m+1)*Gamma[(m+1)/n,-b*x^n*Log[f]]/(n*(-b*x^n*Log[f])^((m+1)/n)) /;
FreeQ[{a,b,f,m,n},x] &&
  NonzeroQ[m+1] &&
  NonzeroQ[m-n+1] &&
  Not[m===-1/2 && ZeroQ[n-1]] &&
  Not[IntegersQ[m,n] && n>0 && (m<-1 || m>=n)] &&
  Not[RationalQ[{m,n}] && (FractionQ[m] || FractionOrNegativeQ[n])]
```

$$\int f^{a+b\,x+c\,x^2}\,dx$$

- **Derivation: Algebraic expansion**
- **Basis:** $a+b\,x+c\,x^2 = \frac{4\,a\,c-b^2}{4\,c} + \frac{(b+2\,c\,x)^2}{4\,c}$
- **Basis:** $f^{z+w} = f^z\,f^w$
- **Rule:**

$$\int f^{a+b\,x+c\,x^2}\,dx \longrightarrow f^{\frac{4\,a\,c-b^2}{4\,c}}\int f^{\frac{(b+2\,c\,x)^2}{4\,c}}\,dx$$

- **Program code:**

```
Int[f_^(a_.+b_.*x_+c_.*x_^2),x_Symbol] :=
  f^(a-b^2/(4*c))*Int[f^((b+2*c*x)^2/(4*c)),x] /;
FreeQ[{a,b,c,f},x]
```

$$\int (d+e\,x)^m\,f^{a+b\,x+c\,x^2}\,dx$$

- **Derivation: Inverted integration by parts**
- **Rule: If** $b\,e-2\,c\,d\neq 0$**, then**

$$\int (d+e\,x)\,f^{a+b\,x+c\,x^2}\,dx\;\longrightarrow\;\frac{e\,f^{a+b\,x+c\,x^2}}{2\,c\,\text{Log}[f]}-\frac{b\,e-2\,c\,d}{2\,c}\int f^{a+b\,x+c\,x^2}\,dx$$

- **Program code:**

```
Int[(d_.+e_.*x_)*f_^(a_.+b_.*x_+c_.*x_^2),x_Symbol] :=
  e*f^(a+b*x+c*x^2)/(2*c*Log[f]) -
  Dist[(b*e-2*c*d)/(2*c),Int[f^(a+b*x+c*x^2),x]] /;
FreeQ[{a,b,c,d,e,f},x] && NonzeroQ[b*e-2*c*d]
```

- **Derivation: Inverted integration by parts**
- **Rule: If** $m>1\,\land\,b\,e-2\,c\,d=0$**, then**

$$\int (d+e\,x)^m\,f^{a+b\,x+c\,x^2}\,dx\;\longrightarrow\;\frac{e\,(d+e\,x)^{m-1}\,f^{a+b\,x+c\,x^2}}{2\,c\,\text{Log}[f]}-\frac{(m-1)\,e^2}{2\,c\,\text{Log}[f]}\int (d+e\,x)^{m-2}\,f^{a+b\,x+c\,x^2}\,dx$$

- **Program code:**

```
Int[(d_.+e_.*x_)^m_*f_^(a_.+b_.*x_+c_.*x_^2),x_Symbol] :=
  e*(d+e*x)^(m-1)*f^(a+b*x+c*x^2)/(2*c*Log[f]) -
  Dist[(m-1)*e^2/(2*c*Log[f]),Int[(d+e*x)^(m-2)*f^(a+b*x+c*x^2),x]] /;
FreeQ[{a,b,c,d,e,f},x] && RationalQ[m] && m>1 && ZeroQ[b*e-2*c*d]
```

- **Derivation: Inverted integration by parts**
- **Rule: If** $m>1\,\land\,b\,e-2\,c\,d\neq 0$**, then**

$$\int (d+e\,x)^m\,f^{a+b\,x+c\,x^2}\,dx\;\longrightarrow\;\frac{e\,(d+e\,x)^{m-1}\,f^{a+b\,x+c\,x^2}}{2\,c\,\text{Log}[f]}-\frac{b\,e-2\,c\,d}{2\,c}\int (d+e\,x)^{m-1}\,f^{a+b\,x+c\,x^2}\,dx-\frac{(m-1)\,e^2}{2\,c\,\text{Log}[f]}\int (d+e\,x)^{m-2}\,f^{a+b\,x+c\,x^2}\,dx$$

- **Program code:**

```
Int[(d_.+e_.*x_)^m_*f_^(a_.+b_.*x_+c_.*x_^2),x_Symbol] :=
  e*(d+e*x)^(m-1)*f^(a+b*x+c*x^2)/(2*c*Log[f]) -
  Dist[(b*e-2*c*d)/(2*c),Int[(d+e*x)^(m-1)*f^(a+b*x+c*x^2),x]] -
  Dist[(m-1)*e^2/(2*c*Log[f]),Int[(d+e*x)^(m-2)*f^(a+b*x+c*x^2),x]] /;
FreeQ[{a,b,c,d,e,f},x] && RationalQ[m] && m>1 && NonzeroQ[b*e-2*c*d]
```

- **Derivation: Integration by parts**
- **Rule: If $m < -1 \land b e - 2 c d = 0$, then**

$$\int (d+e x)^m f^{a+b x+c x^2}\,dx \longrightarrow \frac{(d+e x)^{m+1} f^{a+b x+c x^2}}{e (m+1)} - \frac{2 c \operatorname{Log}[f]}{e^2 (m+1)} \int (d+e x)^{m+2} f^{a+b x+c x^2}\,dx$$

- **Program code:**

```
Int[(d_.+e_.*x_)^m_*f_^(a_.+b_.*x_+c_.*x_^2),x_Symbol] :=
  (d+e*x)^(m+1)*f^(a+b*x+c*x^2)/(e*(m+1)) -
  Dist[2*c*Log[f]/(e^2*(m+1)),Int[(d+e*x)^(m+2)*f^(a+b*x+c*x^2),x]] /;
FreeQ[{a,b,c,d,e,f},x] && RationalQ[m] && m<-1 && ZeroQ[b*e-2*c*d]
```

- **Derivation: Integration by parts**
- **Rule: If $m < -1 \land b e - 2 c d \neq 0$, then**

$$\int (d+e x)^m f^{a+b x+c x^2}\,dx \longrightarrow \frac{(d+e x)^{m+1} f^{a+b x+c x^2}}{e (m+1)} - \frac{(b e - 2 c d) \operatorname{Log}[f]}{e^2 (m+1)} \int (d+e x)^{m+1} f^{a+b x+c x^2}\,dx - \frac{2 c \operatorname{Log}[f]}{e^2 (m+1)} \int (d+e x)^{m+2} f^{a+b x+c x^2}\,dx$$

- **Program code:**

```
Int[(d_.+e_.*x_)^m_*f_^(a_.+b_.*x_+c_.*x_^2),x_Symbol] :=
  (d+e*x)^(m+1)*f^(a+b*x+c*x^2)/(e*(m+1)) -
  Dist[(b*e-2*c*d)*Log[f]/(e^2*(m+1)),Int[(d+e*x)^(m+1)*f^(a+b*x+c*x^2),x]] -
  Dist[2*c*Log[f]/(e^2*(m+1)),Int[(d+e*x)^(m+2)*f^(a+b*x+c*x^2),x]] /;
FreeQ[{a,b,c,d,e,f},x] && RationalQ[m] && m<-1 && NonzeroQ[b*e-2*c*d]
```

$$\int (a+b\,x)^m\, f^{(c+d\,x)^n}\, dx$$

- **Derivation: Integration by parts**
- **Rule: If** $m<-1\land n\in\mathbb{Z}\land n>1$**, then**

$$\int (a+b\,x)^m\, f^{(c+d\,x)^n}\, dx \longrightarrow \frac{(a+b\,x)^{m+1}\, f^{(c+d\,x)^n}}{b\,(m+1)} - \frac{d\,n\,\text{Log}[f]}{b\,(m+1)} \int (a+b\,x)^{m+1}\, f^{(c+d\,x)^n}\,(c+d\,x)^{n-1}\, dx$$

- **Program code:**

```
Int[(a_.+b_.*x_)^m_*f_^((c_.+d_.*x_)^n_.),x_Symbol] :=
  (a+b*x)^(m+1)*f^((c+d*x)^n)/(b*(m+1)) -
  Dist[d*n*Log[f]/(b*(m+1)),Int[(a+b*x)^(m+1)*f^((c+d*x)^n)*(c+d*x)^(n-1),x]] /;
FreeQ[{a,b,c,d},x] && RationalQ[m] && m<-1 && IntegerQ[n] && n>1
```

$$\int \left(a + b\, f^{c+d\,x}\right)^n \,dx$$

- **Reference: CRC 256**
- **Rule: If $d < 0$, then**

$$\int \frac{1}{a + b\, f^{c+d\,x}} \,dx \longrightarrow -\frac{\text{Log}\left[b + a\, f^{-c-d\,x}\right]}{a\, d\, \text{Log}[f]}$$

- **Program code:**

```
Int[1/(a_+b_.*f_^(c_.+d_.*x_)),x_Symbol] :=
  -Log[b+a*f^(-c-d*x)]/(a*d*Log[f]) /;
FreeQ[{a,b,c,d,f},x] && NegativeCoefficientQ[d]
```

- **Reference: CRC 256**
- **Rule: If $\neg\,(d < 0)$, then**

$$\int \frac{1}{a + b\, f^{c+d\,x}} \,dx \longrightarrow \frac{x}{a} - \frac{\text{Log}\left[a + b\, f^{c+d\,x}\right]}{a\, d\, \text{Log}[f]}$$

- **Program code:**

```
Int[1/(a_+b_.*f_^(c_.+d_.*x_)),x_Symbol] :=
  x/a-Log[a+b*f^(c+d*x)]/(a*d*Log[f]) /;
FreeQ[{a,b,c,d,f},x]
```

- **Rule: If $a > 0$, then**

$$\int \frac{1}{\sqrt{a + b\, f^{c+d\,x}}} \,dx \longrightarrow -\frac{2}{\sqrt{a}\, d\, \text{Log}[f]}\, \text{ArcTanh}\left[\frac{\sqrt{a + b\, f^{c+d\,x}}}{\sqrt{a}}\right]$$

- **Program code:**

```
Int[1/Sqrt[a_+b_.*f_^(c_.+d_.*x_)],x_Symbol] :=
  -2*ArcTanh[Sqrt[a+b*f^(c+d*x)]/Sqrt[a]]/(Sqrt[a]*d*Log[f]) /;
FreeQ[{a,b,c,d,f},x] && PosQ[a]
```

- **Rule: If $\neg\,(a > 0)$, then**

$$\int \frac{1}{\sqrt{a + b\, f^{c+d\,x}}}\, dx \;\longrightarrow\; \frac{2}{\sqrt{-a}\; d\,\mathrm{Log}[f]}\,\mathrm{ArcTan}\left[\frac{\sqrt{a + b\, f^{c+d\,x}}}{\sqrt{-a}}\right]$$

- **Program code:**

```
Int[1/Sqrt[a_+b_.*f_^(c_.+d_.*x_)],x_Symbol] :=
  2*ArcTan[Sqrt[a+b*f^(c+d*x)]/Sqrt[-a]]/(Sqrt[-a]*d*Log[f]) /;
FreeQ[{a,b,c,d,f},x] && NegQ[a]
```

- **Rule: If $n > 0$, then**

$$\int \left(a + b\, f^{c+d\,x}\right)^n dx \;\longrightarrow\; \frac{\left(a + b\, f^{c+d\,x}\right)^n}{n\,d\,\mathrm{Log}[f]} + a\int \left(a + b\, f^{c+d\,x}\right)^{n-1} dx$$

- **Program code:**

```
Int[(a_+b_.*f_^(c_.+d_.*x_))^n_,x_Symbol] :=
  (a+b*f^(c+d*x))^n/(n*d*Log[f]) +
  Dist[a,Int[(a+b*f^(c+d*x))^(n-1),x]] /;
FreeQ[{a,b,c,d,f},x] && FractionQ[n] && n>0
```

- **Rule: If $n < -1$, then**

$$\int \left(a + b\, f^{c+d\,x}\right)^n dx \;\longrightarrow\; \frac{\left(a + b\, f^{c+d\,x}\right)^{n+1}}{(n+1)\,a\,d\,\mathrm{Log}[f]} + \frac{1}{a}\int \left(a + b\, f^{c+d\,x}\right)^{n+1} dx$$

- **Program code:**

```
Int[(a_+b_.*f_^(c_.+d_.*x_))^n_,x_Symbol] :=
  -(a+b*f^(c+d*x))^(n+1)/((n+1)*a*d*Log[f]) +
  Dist[1/a,Int[(a+b*f^(c+d*x))^(n+1),x]] /;
FreeQ[{a,b,c,d,f},x] && RationalQ[n] && n<-1
```

$$\int x^m \left(a+b\,f^{c+d\,x}\right)^n dx$$

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{1}{a+b\,z} = \frac{1}{a} - \frac{b\,z}{a\,(a+b\,z)}$
- **Rule: If** $m>0$**, then**

$$\int \frac{x^m}{a+b\,f^{c+d\,x}}\,dx \longrightarrow \frac{x^{m+1}}{a\,(m+1)} - \frac{b}{a}\int \frac{x^m\,f^{c+d\,x}}{a+b\,f^{c+d\,x}}\,dx$$

- **Program code:**

```
Int[x_^m_./(a_+b_.*f_^(c_.+d_.*x_)), x_Symbol] :=
  x^(m+1)/(a*(m+1)) -
  Dist[b/a,Int[x^m*f^(c+d*x)/(a+b*f^(c+d*x)),x]] /;
FreeQ[{a,b,c,d,f},x] && RationalQ[m] && m>0
```

- **Derivation: Integration by parts**
- **Rule: If** $m>0 \land n<-1$**, then**

$$\int x^m \left(a+b\,f^{c+d\,x}\right)^n dx \longrightarrow x^m \int \left(a+b\,f^{c+d\,x}\right)^n dx - m\int \left(x^{m-1}\int \left(a+b\,f^{c+d\,x}\right)^n dx\right) dx$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*f_^(c_.+d_.*x_))^n_, x_Symbol] :=
  Module[{u=Block[{ShowSteps=False,StepCounter=Null}, Int[(a+b*f^(c+d*x))^n,x]]},
  x^m*u - Dist[m,Int[x^(m-1)*u,x]]] /;
FreeQ[{a,b,c,d,f},x] && RationalQ[{m,n}] && m>0 && n<-1
```

$$\int x^m f^{c\,(a+b\,x)^n}\,dx$$

- **Rule: If $m>1$, then**

$$\int x^m f^{c\,(a+b\,x)^2}\,dx \longrightarrow \int x^m f^{a^2\,c+2\,a\,b\,c\,x+b^2\,c\,x^2}\,dx$$

- **Program code:**

```
Int[x_^m_*f_^(c_.*(a_+b_.*x_)^2),x_Symbol] :=
  Int[x^m*f^(a^2*c+2*a*b*c*x+b^2*c*x^2),x] /;
FreeQ[{a,b,c,f},x] && FractionQ[m] && m>1
```

- **Rule: If $m\in\mathbb{Z}\wedge m>0$, then**

$$\int x^m \left(a+b\,f^{c+d\,x}\right)^n dx \longrightarrow \frac{1}{b^m}\int b^m\,x^m-(a+b\,x)^m\,f^{c\,(a+b\,x)^n}\,dx+\frac{1}{b^{m+1}}\,\text{Subst}\left[\int x^m\,f^{c\,x^n}\,dx,\;x,\;a+b\,x\right]$$

- **Program code:**

```
Int[x_^m_.*f_^(c_.*(a_+b_.*x_)^n_),x_Symbol] :=
  Dist[1/b^m,Int[Expand[b^m*x^m-(a+b*x)^m,x]*f^(c*(a+b*x)^n),x]] +
  Dist[1/b^(m+1),Subst[Int[x^m*f^(c*x^n),x],x,a+b*x]] /;
FreeQ[{a,b,c,f,n},x] && IntegerQ[m] && m>0
```

$$\int x^m f^{c+d\,x}\left(a+b\,f^{e+f\,x}\right)^n\,dx$$

- **Derivation: Integration by parts**
- **Rule: If** $m>1$, **then**

$$\int\frac{x^m\,f^{c+d\,x}}{a+b\,f^{c+d\,x}}\,dx\;\longrightarrow\;\frac{x^m\,\text{Log}\left[1+\frac{b\,f^{c+d\,x}}{a}\right]}{b\,d\,\text{Log}[f]}-\frac{m}{b\,d\,\text{Log}[f]}\int x^{m-1}\,\text{Log}\left[1+\frac{b\,f^{c+d\,x}}{a}\right]dx$$

- **Program code:**

```
Int[x_^m_.*f_^(c_.+d_.*x_)/(a_+b_.*f_^(c_.+d_.*x_)), x_Symbol] :=
  x^m*Log[1+b*f^(c+d*x)/a]/(b*d*Log[f]) -
  Dist[m/(b*d*Log[f]),Int[x^(m-1)*Log[1+b/a*f^(c+d*x)],x]] /;
FreeQ[{a,b,c,d,f},x] && RationalQ[m] && m>=1
```

- **Derivation: Integration by parts**
- **Rule: If** $m>0\,\wedge\,n\in\mathbb{Z}\,\wedge\,n<0$, **then**

$$\int x^m\,f^{c+d\,x}\left(a+b\,f^{2\,(c+d\,x)}\right)^n dx\;\longrightarrow$$
$$x^m\int f^{c+d\,x}\left(a+b\,f^{2\,(c+d\,x)}\right)^n dx-m\int x^{m-1}\left(\int f^{c+d\,x}\left(a+b\,f^{2\,(c+d\,x)}\right)^n dx\right)dx$$

- **Program code:**

```
Int[x_^m_.*f_^(c_.+d_.*x_)*(a_.+b_.*f_^v_)^n_,x_Symbol] :=
  Module[{u=Block[{ShowSteps=False,StepCounter=Null}, Int[f^(c+d*x)*(a+b*f^v)^n,x]]},
  x^m*u - Dist[m,Int[x^(m-1)*u,x]]] /;
FreeQ[{a,b,c,d,f},x] && ZeroQ[2*(c+d*x)-v] && RationalQ[m] && m>0 && IntegerQ[n] && n<0
```

- **Derivation: Integration by parts**
- **Rule: If** $m>0$, **then**

$$\int\frac{x^m}{a\,f^{c+d\,x}+b\,f^{-(c+d*x)}}\,dx\;\longrightarrow\;x^m\int\frac{1}{a\,f^{c+d\,x}+b\,f^{-(c+d*x)}}\,dx-m\int x^{m-1}\int\frac{1}{a\,f^{c+d\,x}+b\,f^{-(c+d*x)}}\,dx\,dx$$

- **Program code:**

```
Int[x_^m_./(a_.*f_^(c_.+d_.*x_)+b_.*f_^v_),x_Symbol] :=
  Module[{u=Block[{ShowSteps=False,StepCounter=Null}, Int[1/(a*f^(c+d*x)+b*f^v),x]]},
  x^m*u - Dist[m,Int[x^(m-1)*u,x]]] /;
FreeQ[{a,b,c,d,f},x] && ZeroQ[(c+d*x)+v] && RationalQ[m] && m>0
```

$$\int (a+b\,x)^m\,f^{e\,(c+d\,x)^n}\,dx$$

- **Derivation: Integration by substitution**
- **Rule: If $p \in \mathbb{Q}$, then**

$$\int \frac{x^m\,f^{c+d\,x}}{a+b\,f^{c+d\,x}}\,dx \longrightarrow \frac{1}{b}\,\text{Subst}\Big[\int x^m\left(f^{e\left(c-\frac{a\,d}{b}+\frac{d\,x}{b}\right)^n}\right)^p dx,\;x,\;a+b\,x\Big]$$

- **Program code:**

```
Int[(a_.+b_.*x_)^m_.*(f_^(e_.*(c_.+d_.*x_)^n_.))^p_.,x_Symbol] :=
  Dist[1/b,Subst[Int[x^m*(f^(e*(c-a*d/b+d*x/b)^n))^p,x],x,a+b*x]] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && RationalQ[p] && Not[a===0 && b===1]
```

$$\int f^{\frac{a+b\,x^4}{x^2}}\,dx$$

- **Derivation: Integration by substitution**

- **Rule: If $p \in \mathbb{Q}$, then**

$$\int f^{\frac{a+b\,x^4}{x^2}}\,dx \to \frac{\sqrt{\pi}\ \mathrm{Exp}\left[2\sqrt{-a\,\mathrm{Log}[f]}\ \sqrt{-b\,\mathrm{Log}[f]}\right]\ \mathrm{Erf}\left[\frac{\sqrt{-a\,\mathrm{Log}[f]}+\sqrt{-b\,\mathrm{Log}[f]}\ x^2}{x}\right]}{4\sqrt{-b\,\mathrm{Log}[f]}} - \frac{\sqrt{\pi}\ \mathrm{Exp}\left[-2\sqrt{-a\,\mathrm{Log}[f]}\ \sqrt{-b\,\mathrm{Log}[f]}\right]\ \mathrm{Erf}\left[\frac{\sqrt{-a\,\mathrm{Log}[f]}-\sqrt{-b\,\mathrm{Log}[f]}\ x^2}{x}\right]}{4\sqrt{-b\,\mathrm{Log}[f]}}$$

- **Program code:**

```
Int[f_^((a_.+b_.*x_^4)/x_^2),x_Symbol] :=
  Sqrt[Pi]*Exp[2*Sqrt[-a*Log[f]]*Sqrt[-b*Log[f]]]*Erf[(Sqrt[-a*Log[f]]+Sqrt[-b*Log[f]]*x^2)/x]/
    (4*Sqrt[-b*Log[f]]) -
  Sqrt[Pi]*Exp[-2*Sqrt[-a*Log[f]]*Sqrt[-b*Log[f]]]*Erf[(Sqrt[-a*Log[f]]-Sqrt[-b*Log[f]]*x^2)/x]/
    (4*Sqrt[-b*Log[f]]) /;
FreeQ[{a,b,f},x]
```

$$\int \frac{u}{a + b f^{d+e x} + c f^{g+h x}} \, dx$$

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{1}{a+b f^z+c f^{2 z}} = \frac{1}{a} - \frac{f^z (b+c f^z)}{a \left(a+b f^z+c f^{2 z}\right)}$
- **Rule:**

$$\int \frac{1}{a + b f^{d+e x} + c f^{2 (d+e x)}} \, dx \longrightarrow \frac{x}{a} - \frac{1}{a} \int \frac{f^{d+e x} \left(b + c f^{d+e x}\right)}{a + b f^{d+e x} + c f^{2 (d+e x)}} \, dx$$

- **Program code:**

```
Int[1/(a_+b_.*f_^u_+c_.*f_^v_), x_Symbol] :=
  x/a -
  Dist[1/a,Int[f^u*(b+c*f^u)/(a+b*f^u+c*f^v),x]] /;
FreeQ[{a,b,c,f},x] && LinearQ[u,x] && LinearQ[v,x] && ZeroQ[2*u-v] &&
Not[RationalQ[Rt[b^2-4*a*c,2]]]
```

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{d+e f^z}{a+b f^z+c f^{2 z}} = \frac{d}{a} - \frac{f^z (b d-a e+c d f^z)}{a \left(a+b f^z+c f^{2 z}\right)}$
- **Rule:**

$$\int \frac{d + e f^{d+e x}}{a + b f^{d+e x} + c f^{2 (d+e x)}} \, dx \longrightarrow \frac{d x}{a} - \frac{1}{a} \int \frac{f^{d+e x} \left(b d - a e + c d f^{d+e x}\right)}{a + b f^{d+e x} + c f^{2 (d+e x)}} \, dx$$

- **Program code:**

```
Int[(d_+e_.*f_^u_)/(a_+b_.*f_^u_+c_.*f_^v_), x_Symbol] :=
  d*x/a -
  Dist[1/a,Int[f^u*(b*d-a*e+c*d*f^u)/(a+b*f^u+c*f^v),x]] /;
FreeQ[{a,b,c,d,e,f},x] && LinearQ[u,x] && LinearQ[v,x] && ZeroQ[2*u-v] &&
Not[RationalQ[Rt[b^2-4*a*c,2]]]
```

- **Derivation: Algebraic simplification**
- **Basis:** $\frac{1}{a+b\,z+\frac{c}{z}} = \frac{z}{c+a\,z+b\,z^2}$
- **Rule:**

$$\int \frac{u}{a+b\,f^{d+e\,x}+c\,f^{-(d+e\,x)}}\,dx \;\longrightarrow\; \int \frac{u\,f^{d+e\,x}}{c+a\,f^{d+e\,x}+b\,f^{2\,(d+e\,x)}}\,dx$$

- **Program code:**

```
Int[u_/(a_+b_.*f_^v_+c_.*f_^w_), x_Symbol] :=
  Int[u*f^v/(c+a*f^v+b*f^(2*v)),x] /;
FreeQ[{a,b,c,f},x] && LinearQ[v,x] && LinearQ[w,x] && ZeroQ[v+w] &&
If[RationalQ[Coefficient[v,x,1]], Coefficient[v,x,1]>0, LeafCount[v]<LeafCount[w]]
```

$$\int x^m \left(e^x + x^m\right)^n dx$$

- **Derivation: Algebraic expansion**

- **Basis:** $\int f[x] \left(e^x + f[x]\right)^n dx == -\frac{(e^x+f[x])^{n+1}}{n+1} + \int (e^x + f[x])^{n+1} dx + \int f'[x] \left(e^x + f[x]\right)^n dx$

- **Rule: If** $m > 0 \wedge n + 1 \neq 0$**, then**

$$\int x^m \left(e^x + x^m\right)^n dx \longrightarrow -\frac{\left(e^x + x^m\right)^{n+1}}{n+1} + \int \left(e^x + x^m\right)^{n+1} dx + m \int x^{m-1} \left(e^x + x^m\right)^n dx$$

- **Program code:**

```
Int[x_^m_.*(E^x_+x_^m_.)^n_,x_Symbol] :=
  -(E^x+x^m)^(n+1)/(n+1) +
  Int[(E^x+x^m)^(n+1),x] +
  Dist[m,Int[x^(m-1)*(E^x+x^m)^n,x]] /;
RationalQ[{m,n}] && m>0 && NonzeroQ[n+1]
```