$$\int x^m\,(a+b\,\text{Sin}[c+d\,x])^n\,dx$$

- **Derivation: Algebraic simplification**
- **Basis: If** $a^2-b^2=0$, **then** $a+b\,\text{Sin}[z]=2\,a\,\text{Cos}\left[-\frac{\pi a}{4\,b}+\frac{z}{2}\right]^2$
- **Rule: If** $a^2-b^2=0\,\land\,m\in\mathbb{Q}\,\land\,n\in\mathbb{Z}\,\land\,n<0$, **then**

$$\int x^m\,(a+b\,\text{Sin}[c+d\,x])^n\,dx\;\longrightarrow\;(2\,a)^n\int x^m\,\text{Cos}\left[-\frac{\pi a}{4\,b}+\frac{c}{2}+\frac{d\,x}{2}\right]^{2\,n}dx$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*Sin[c_.+d_.*x_])^n_,x_Symbol] :=
  Dist[(2*a)^n,Int[x^m*Cos[-Pi*a/(4*b)+c/2+d*x/2]^(2*n),x]] /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2-b^2] && RationalQ[m] && IntegerQ[n] && n<0
```

- **Derivation: Algebraic simplification and piecewise constant extraction**
- **Basis: If** $a^2-b^2=0$, **then** $a+b\,\text{Sin}[z]=2\,a\,\text{Cos}\left[-\frac{\pi a}{4\,b}+\frac{z}{2}\right]^2$
- **Basis: If** $a^2-b^2=0$, **then** $\partial_z\frac{\sqrt{a+b\,\text{Sin}[z]}}{\text{Cos}\left[-\frac{\pi a}{4\,b}+\frac{z}{2}\right]}=0$
- **Rule: If** $a^2-b^2=0\,\land\,m\in\mathbb{Q}\,\land\,n-\frac{1}{2}\in\mathbb{Z}$, **then**

$$\int x^m\,(a+b\,\text{Sin}[c+d\,x])^n\,dx\;\longrightarrow\;\frac{(2\,a)^{n-\frac{1}{2}}\sqrt{a+b\,\text{Sin}[c+d\,x]}}{\text{Cos}\left[-\frac{\pi a}{4\,b}+\frac{c}{2}+\frac{d\,x}{2}\right]}\int x^m\,\text{Cos}\left[-\frac{\pi a}{4\,b}+\frac{c}{2}+\frac{d\,x}{2}\right]^{2\,n}dx$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*Sin[c_.+d_.*x_])^n_,x_Symbol] :=
  Dist[(2*a)^(n-1/2)*Sqrt[a+b*Sin[c+d*x]]/Cos[-Pi*a/(4*b)+c/2+d*x/2],
    Int[x^m*Cos[-Pi*a/(4*b)+c/2+d*x/2]^(2*n),x]] /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2-b^2] && RationalQ[m] && IntegerQ[n-1/2]
```

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{1}{(a+b z)^2} = \frac{a}{\left(a^2-b^2\right) (a+b z)} - \frac{b (b+a z)}{\left(a^2-b^2\right) (a+b z)^2}$
- **Rule: If** $a^2 - b^2 \neq 0$**, then**

$$\int \frac{x}{(a + b \operatorname{Sin}[c + d x])^2} \, dx \longrightarrow \frac{a}{a^2 - b^2} \int \frac{x}{a + b \operatorname{Sin}[c + d x]} \, dx - \frac{b}{a^2 - b^2} \int \frac{x (b + a \operatorname{Sin}[c + d x])}{(a + b \operatorname{Sin}[c + d x])^2} \, dx$$

- **Program code:**

```
Int[x_/(a_+b_.*Sin[c_.+d_.*x_])^2,x_Symbol] :=
  Dist[a/(a^2-b^2),Int[x/(a+b*Sin[c+d*x]),x]] -
  Dist[b/(a^2-b^2),Int[x*(b+a*Sin[c+d*x])/(a+b*Sin[c+d*x])^2,x]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2-b^2]
```

- **Derivation: Algebraic expansion**
- **Basis:** $a + b \operatorname{Sin}[z] = \frac{i b+2 a e^{i z}-i b e^{2 i z}}{2 e^{i z}}$
- **Rule: If** $a^2 - b^2 \neq 0 \land m > 0 \land n \in \mathbb{Z} \land n < 0$**, then**

$$\int x^m (a + b \operatorname{Sin}[c + d x])^n \, dx \longrightarrow \frac{1}{2^n} \int \frac{x^m \left(i b + 2 a e^{i c+i d x} - i b e^{2 (i c+i d x)}\right)^n}{e^{n (i c+i d x)}} \, dx$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*Sin[c_.+d_.*x_])^n_,x_Symbol] :=
  Dist[1/2^n,Int[x^m*(I*b+2*a*E^(I*c+I*d*x)-I*b*E^(2*(I*c+I*d*x)))^n/E^(n*(I*c+I*d*x)),x]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2-b^2] && RationalQ[m] && m>0 && IntegerQ[n] && n<0
```

$$\int x^m (a+b\,\text{Cos}[c+d\,x])^n\,dx$$

- **Derivation: Algebraic simplification**
- **Basis: If** $a^2-b^2=0$, **then** $a+b\,\text{Cos}[z]=2\,a\,\text{Cos}\left[-\frac{1}{4}\pi\left(1-\frac{a}{b}\right)+\frac{z}{2}\right]^2$
- **Note: This rule unifies the following two rules, but superficially appears more complicated.**
- **Rule: If** $a^2-b^2=0\bigwedge m\in\mathbb{Q}\bigwedge n\in\mathbb{Z}\bigwedge n<0$, **then**

$$\int x^m (a+b\,\text{Cos}[c+d\,x])^n\,dx \longrightarrow (2\,a)^n\int x^m\,\text{Cos}\left[\frac{1}{4}(-\pi)\left(1-\frac{a}{b}\right)+\frac{c}{2}+\frac{d\,x}{2}\right]^{2\,n}dx$$

- **Program code:**

```
(* Int[x_^m_.*(a_+b_.*Cos[c_.+d_.*x_])^n_,x_Symbol] :=
  Dist[(2*a)^n,Int[x^m*Cos[-Pi/4*(1-a/b)+c/2+d*x/2]^(2*n),x]] /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2-b^2] && RationalQ[m] && IntegerQ[n] && n<0 *)
```

- **Derivation: Algebraic simplification**
- **Basis:** $1+\text{Cos}[z]=2\,\text{Cos}\left[\frac{z}{2}\right]^2$
- **Rule: If** $a-b=0\bigwedge m\in\mathbb{Q}\bigwedge n\in\mathbb{Z}\bigwedge n<0$, **then**

$$\int x^m (a+b\,\text{Cos}[c+d\,x])^n\,dx \longrightarrow (2\,a)^n\int x^m\,\text{Cos}\left[\frac{c}{2}+\frac{d\,x}{2}\right]^{2\,n}dx$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*Cos[c_.+d_.*x_])^n_,x_Symbol] :=
  Dist[(2*a)^n,Int[x^m*Cos[c/2+d*x/2]^(2*n),x]] /;
FreeQ[{a,b,c,d},x] && ZeroQ[a-b] && RationalQ[m] && IntegerQ[n] && n<0
```

- **Derivation: Algebraic simplification**
- **Basis:** $1 - \text{Cos}[z] = 2\,\text{Sin}\left[\frac{z}{2}\right]^2$
- **Rule: If** $a + b = 0 \bigwedge m \in \mathbb{Q} \bigwedge n \in \mathbb{Z} \bigwedge n < 0$**, then**

$$\int x^m\,(a + b\,\text{Cos}[c + d\,x])^n\,dx \longrightarrow (2\,a)^n \int x^m\,\text{Sin}\left[\frac{c}{2} + \frac{d\,x}{2}\right]^{2\,n}dx$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*Cos[c_.+d_.*x_])^n_,x_Symbol] :=
  Dist[(2*a)^n,Int[x^m*Sin[c/2+d*x/2]^(2*n),x]] /;
FreeQ[{a,b,c,d},x] && ZeroQ[a+b] && RationalQ[m] && IntegerQ[n] && n<0
```

- **Derivation: Algebraic simplification and piecewise constant extraction**
- **Basis: If** $a^2 - b^2 = 0$**, then** $a + b\,\text{Cos}[z] = 2\,a\,\text{Cos}\left[\frac{z}{2} - \frac{1}{4}\,\pi\,\left(1 - \frac{a}{b}\right)\right]^2$
- **Basis: If** $a^2 - b^2 = 0$**, then** $\partial_z \frac{\sqrt{a+b\,\text{Cos}[z]}}{\text{Cos}\left[\frac{z}{2}-\frac{1}{4}\,\pi\,\left(1-\frac{a}{b}\right)\right]} == 0$
- **Note: This rule unifies the following two rules, but superficially appears more complicated.**
- **Rule: If** $a^2 - b^2 = 0 \bigwedge m \in \mathbb{Q} \bigwedge n - \frac{1}{2} \in \mathbb{Z}$**, then**

$$\int x^m\,(a + b\,\text{Cos}[c + d\,x])^n\,dx \longrightarrow \frac{(2\,a)^{n-\frac{1}{2}}\,\sqrt{a + b\,\text{Cos}[c + d\,x]}}{\text{Cos}\left[\frac{1}{4}\,(-\pi)\,\left(1 - \frac{a}{b}\right) + \frac{c}{2} + \frac{d\,x}{2}\right]} \int x^m\,\text{Cos}\left[\frac{1}{4}\,(-\pi)\,\left(1 - \frac{a}{b}\right) + \frac{c}{2} + \frac{d\,x}{2}\right]^{2\,n}dx$$

- **Program code:**

```
(* Int[x_^m_.*(a_+b_.*Cos[c_.+d_.*x_])^n_,x_Symbol] :=
  Dist[(2*a)^(n-1/2)*Sqrt[a+b*Cos[c+d*x]]/Cos[-Pi/4*(1-a/b)+c/2+d*x/2],
    Int[x^m*Cos[-Pi/4*(1-a/b)+c/2+d*x/2]^(2*n),x]] /;
FreeQ[{a,b,c,d},x] && ZeroQ[a^2-b^2] && RationalQ[m] && IntegerQ[n-1/2] *)
```

- **Derivation: Algebraic simplification**
- **Basis:** $1+\text{Cos}[z]=2\,\text{Cos}\left[\frac{z}{2}\right]^2$
- **Basis:** $\partial_z \frac{\sqrt{a+a\,\text{Cos}[z]}}{\text{Cos}\left[\frac{z}{2}\right]}=0$
- **Rule: If** $a-b=0\ \bigwedge\ m\in\mathbb{Q}\ \bigwedge\ n-\frac{1}{2}\in\mathbb{Z}$, **then**

$$\int x^m\,(a+b\,\text{Cos}[c+d\,x])^n\,dx\ \longrightarrow\ \frac{(2\,a)^{n-\frac{1}{2}}\sqrt{a+b\,\text{Cos}[c+d\,x]}}{\text{Cos}\left[\frac{c}{2}+\frac{d\,x}{2}\right]}\int x^m\,\text{Cos}\left[\frac{c}{2}+\frac{d\,x}{2}\right]^{2\,n}\,dx$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*Cos[c_.+d_.*x_])^n_,x_Symbol] :=
  Dist[(2*a)^(n-1/2)*Sqrt[a+b*Cos[c+d*x]]/Cos[c/2+d*x/2],Int[x^m*Cos[c/2+d*x/2]^(2*n),x]] /;
FreeQ[{a,b,c,d},x] && ZeroQ[a-b] && RationalQ[m] && IntegerQ[n-1/2]
```

- **Derivation: Algebraic simplification**
- **Basis:** $1-\text{Cos}[z]=2\,\text{Sin}\left[\frac{z}{2}\right]^2$
- **Basis:** $\partial_z \frac{\sqrt{a-a\,\text{Cos}[z]}}{\text{Sin}\left[\frac{z}{2}\right]}=0$
- **Rule: If** $a+b=0\ \bigwedge\ m\in\mathbb{Q}\ \bigwedge\ n-\frac{1}{2}\in\mathbb{Z}$, **then**

$$\int x^m\,(a+b\,\text{Cos}[c+d\,x])^n\,dx\ \longrightarrow\ \frac{(2\,a)^{n-\frac{1}{2}}\sqrt{a+b\,\text{Cos}[c+d\,x]}}{\text{Sin}\left[\frac{c}{2}+\frac{d\,x}{2}\right]}\int x^m\,\text{Sin}\left[\frac{c}{2}+\frac{d\,x}{2}\right]^{2\,n}\,dx$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*Cos[c_.+d_.*x_])^n_,x_Symbol] :=
  Dist[(2*a)^(n-1/2)*Sqrt[a+b*Cos[c+d*x]]/Sin[c/2+d*x/2],Int[x^m*Sin[c/2+d*x/2]^(2*n),x]] /;
FreeQ[{a,b,c,d},x] && ZeroQ[a+b] && RationalQ[m] && IntegerQ[n-1/2]
```

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{1}{(a+b z)^2} = \frac{a}{\left(a^2-b^2\right) (a+b z)} - \frac{b (b+a z)}{\left(a^2-b^2\right) (a+b z)^2}$
- **Rule: If** $a^2 - b^2 \neq 0$**, then**

$$\int \frac{x}{(a + b \operatorname{Cos}[c + d x])^2} \, dx \longrightarrow \frac{a}{a^2 - b^2} \int \frac{x}{a + b \operatorname{Cos}[c + d x]} \, dx - \frac{b}{a^2 - b^2} \int \frac{x (b + a \operatorname{Cos}[c + d x])}{(a + b \operatorname{Cos}[c + d x])^2} \, dx$$

- **Program code:**

```
Int[x_/(a_+b_.*Cos[c_.+d_.*x_])^2,x_Symbol] :=
  Dist[a/(a^2-b^2),Int[x/(a+b*Cos[c+d*x]),x]] -
  Dist[b/(a^2-b^2),Int[x*(b+a*Cos[c+d*x])/(a+b*Cos[c+d*x])^2,x]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2-b^2]
```

- **Derivation: Algebraic expansion**
- **Basis:** $a + b \operatorname{Cos}[z] = \frac{b+2 a e^{i z}+b e^{2 i z}}{2 e^{i z}}$
- **Rule: If** $a^2 - b^2 \neq 0 \land m > 0 \land n \in \mathbb{Z} \land n < 0$**, then**

$$\int x^m (a + b \operatorname{Cos}[c + d x])^n \, dx \longrightarrow \frac{1}{2^n} \int \frac{x^m \left(b + 2 a e^{i c+i d x} + b e^{2 (i c+i d x)}\right)^n}{e^{n (i c+i d x)}} \, dx$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*Cos[c_.+d_.*x_])^n_,x_Symbol] :=
  Dist[1/2^n,Int[x^m*(b+2*a*E^(I*c+I*d*x)+b*E^(2*(I*c+I*d*x)))^n/E^(n*(I*c+I*d*x)),x]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2-b^2] && RationalQ[m] && m>0 && IntegerQ[n] && n<0
```

$$\int u \left(a + b \operatorname{Sin}[c + d x]^2\right)^n dx$$

- **Derivation: Algebraic simplification**
- **Basis:** $\operatorname{Sin}[z]^2 = \frac{1}{2} (1 - \operatorname{Cos}[2 z])$
- **Note: This rule should be replaced with rules that directly reduce the integrand rather than transforming it using trig power expansion!**
- **Rule: If** $a + b \neq 0 \wedge n \neq -1$**, then**

$$\int \left(a + b \operatorname{Sin}[c + d x]^2\right)^n dx \longrightarrow \frac{1}{2^n} \int (2 a + b - b \operatorname{Cos}[2 c + 2 d x])^n dx$$

- **Program code:**

```
Int[(a_+b_.*Sin[c_.+d_.*x_]^2)^n_,x_Symbol] :=
  Dist[1/2^n,Int[(2*a+b-b*Cos[2*c+2*d*x])^n,x]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a+b] && RationalQ[n] && n≠-1
```

```
Int[(a_+b_.*Cos[c_.+d_.*x_]^2)^n_,x_Symbol] :=
  Dist[1/2^n,Int[(2*a+b+b*Cos[2*c+2*d*x])^n,x]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a+b] && RationalQ[n] && n≠-1
```

- **Derivation: Algebraic simplification**
- **Basis:** $\operatorname{Sin}[z]^2 = \frac{1}{2} (1 - \operatorname{Cos}[2 z])$
- **Note: This rule should be replaced with rules that directly reduce the integrand rather than transforming it using trig power expansion!**
- **Rule: If** $a + b \neq 0 \wedge m \in \mathbb{Z} \wedge m > 0$**, then**

$$\int x^m \left(a + b \operatorname{Sin}[c + d x]^2\right)^n dx \longrightarrow \frac{1}{2^n} \int x^m (2 a + b - b \operatorname{Cos}[2 c + 2 d x])^n dx$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*Sin[c_.+d_.*x_]^2)^n_,x_Symbol] :=
  Dist[1/2^n,Int[x^m*(2*a+b-b*Cos[2*c+2*d*x])^n,x]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a+b] && IntegersQ[m,n] && (m>0 && n==-1 || m==1 && n==-2)
```

```
Int[x_^m_.*(a_+b_.*Cos[c_.+d_.*x_]^2)^n_,x_Symbol] :=
  Dist[1/2^n,Int[x^m*(2*a+b+b*Cos[2*c+2*d*x])^n,x]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a+b] && IntegersQ[m,n] && (m>0 && n==-1 || m==1 && n==-2)
```

$$\int \mathrm{Sin}[a + b\,x^n]\,dx$$

- **Derivation: Primitive rule**
- **Basis:** $\mathrm{FresnelS}'[z] = \mathrm{Sin}\left[\frac{\pi z^2}{2}\right]$
- **Rule:**

$$\int \mathrm{Sin}\left[b\,x^2\right] dx \longrightarrow \frac{\sqrt{\frac{\pi}{2}}}{\sqrt{b}}\,\mathrm{FresnelS}\left[\frac{\sqrt{b}\,x}{\sqrt{\frac{\pi}{2}}}\right]$$

- **Program code:**

```
Int[Sin[b_.*x_^2],x_Symbol] :=
  Sqrt[Pi/2]*FresnelS[Rt[b,2]*x/Sqrt[Pi/2]]/Rt[b,2] /;
FreeQ[b,x]
```

```
Int[Cos[b_.*x_^2],x_Symbol] :=
  Sqrt[Pi/2]*FresnelC[Rt[b,2]*x/Sqrt[Pi/2]]/Rt[b,2] /;
FreeQ[b,x]
```

- **Derivation: Algebraic expansion**
- **Basis:** $\mathrm{Sin}[w + z] = \mathrm{Sin}[w]\,\mathrm{Cos}[z] + \mathrm{Cos}[w]\,\mathrm{Sin}[z]$
- **Rule:**

$$\int \mathrm{Sin}\left[a + b\,x^2\right] dx \longrightarrow \mathrm{Sin}[a]\int \mathrm{Cos}\left[b\,x^2\right] dx + \mathrm{Cos}[a]\int \mathrm{Sin}\left[b\,x^2\right] dx$$

- **Program code:**

```
Int[Sin[a_+b_.*x_^2],x_Symbol] :=
  Dist[Sin[a],Int[Cos[b*x^2],x]] +
  Dist[Cos[a],Int[Sin[b*x^2],x]] /;
FreeQ[{a,b},x]
```

```
Int[Cos[a_+b_.*x_^2],x_Symbol] :=
  Dist[Cos[a],Int[Cos[b*x^2],x]] -
  Dist[Sin[a],Int[Sin[b*x^2],x]] /;
FreeQ[{a,b},x]
```

- **Derivation: Algebraic expansion**
- **Basis:** $\mathrm{Sin}[z] = \frac{1}{2} i e^{-i z} - \frac{1}{2} i e^{i z}$
- **Rule: If** $\neg (n \in \mathbb{F} \lor n < 0)$**, then**

$$\int \mathrm{Sin}[a + b x^n] \, dx \longrightarrow \frac{i}{2} \int e^{-a i - b i x^n} \, dx - \frac{i}{2} \int e^{a i + b i x^n} \, dx$$

- **Program code:**

```
Int[Sin[a_.+b_.*x_^n_],x_Symbol] :=
  Dist[I/2,Int[E^(-a*I-b*I*x^n),x]] -
  Dist[I/2,Int[E^(a*I+b*I*x^n),x]] /;
FreeQ[{a,b,n},x] && Not[FractionOrNegativeQ[n]]
```

- **Basis:** $\mathrm{Cos}[z] = \frac{1}{2} e^{-i z} + \frac{1}{2} e^{i z}$

```
Int[Cos[a_.+b_.*x_^n_],x_Symbol] :=
  Dist[1/2,Int[E^(-a*I-b*I*x^n),x]] +
  Dist[1/2,Int[E^(a*I+b*I*x^n),x]] /;
FreeQ[{a,b,n},x] && Not[FractionOrNegativeQ[n]]
```

- **Derivation: Integration by parts**
- **Note: Although resulting integrand looks more complicated than the original, rules for improper binomials rectify it.**
- **Rule: If** $n \in \mathbb{Z} \lor n < 0$**, then**

$$\int \mathrm{Sin}[a + b x^n] \, dx \longrightarrow x \, \mathrm{Sin}[a + b x^n] - b n \int x^n \, \mathrm{Cos}[a + b x^n] \, dx$$

- **Program code:**

```
Int[Sin[a_.+b_.*x_^n_],x_Symbol] :=
  x*Sin[a+b*x^n] -
  Dist[b*n,Int[x^n*Cos[a+b*x^n],x]] /;
FreeQ[{a,b},x] && IntegerQ[n] && n<0
```

```
Int[Cos[a_.+b_.*x_^n_],x_Symbol] :=
  x*Cos[a+b*x^n] +
  Dist[b*n,Int[x^n*Sin[a+b*x^n],x]] /;
FreeQ[{a,b},x] && IntegerQ[n] && n<0
```

$$\int x^m \operatorname{Sin}[a + b x^n] \, dx$$

- **Derivation: Primitive rule**
- **Basis:** $\operatorname{SinIntegral}'[z] = \frac{\operatorname{Sin}[z]}{z}$
- **Rule:**

$$\int \frac{\operatorname{Sin}[b x^n]}{x} \, dx \longrightarrow \frac{\operatorname{SinIntegral}[b x^n]}{n}$$

- **Program code:**

```
Int[Sin[b_.*x_^n_.]/x_,x_Symbol] :=
  SinIntegral[b*x^n]/n /;
FreeQ[{b,n},x]
```

```
Int[Cos[b_.*x_^n_.]/x_,x_Symbol] :=
  CosIntegral[b*x^n]/n /;
FreeQ[{b,n},x]
```

- **Derivation: Algebraic expansion**
- **Basis:** $\operatorname{Sin}[w + z] = \operatorname{Sin}[w] \operatorname{Cos}[z] + \operatorname{Cos}[w] \operatorname{Sin}[z]$
- **Rule:**

$$\int \frac{\operatorname{Sin}[a + b x^n]}{x} \, dx \longrightarrow \operatorname{Sin}[a] \int \frac{\operatorname{Cos}[b x^n]}{x} \, dx + \operatorname{Cos}[a] \int \frac{\operatorname{Sin}[b x^n]}{x} \, dx$$

- **Program code:**

```
Int[Sin[a_+b_.*x_^n_.]/x_,x_Symbol] :=
  Dist[Sin[a],Int[Cos[b*x^n]/x,x]] +
  Dist[Cos[a],Int[Sin[b*x^n]/x,x]] /;
FreeQ[{a,b,n},x]
```

```
Int[Cos[a_+b_.*x_^n_.]/x_,x_Symbol] :=
  Dist[Cos[a],Int[Cos[b*x^n]/x,x]] -
  Dist[Sin[a],Int[Sin[b*x^n]/x,x]] /;
FreeQ[{a,b,n},x]
```

- **Reference: CRC 392, A&S 4.3.119**
- **Derivation: Integration by parts**
- **Basis:** $x^m \operatorname{Sin}[a + b x^n] = -\frac{x^{m-n+1} \partial_x \operatorname{Cos}[a+b x^n]}{b n}$
- **Rule: If** $n \in \mathbb{Z} \wedge 0 < n \le m$**, then**

$$\int x^m \operatorname{Sin}[a + b x^n] \, dx \longrightarrow -\frac{x^{m-n+1} \operatorname{Cos}[a + b x^n]}{b n} + \frac{m-n+1}{b n} \int x^{m-n} \operatorname{Cos}[a + b x^n] \, dx$$

- **Program code:**

```
Int[x_^m_.*Sin[a_.+b_.*x_^n_.],x_Symbol] :=
  -x^(m-n+1)*Cos[a+b*x^n]/(b*n) +
  Dist[(m-n+1)/(b*n),Int[x^(m-n)*Cos[a+b*x^n],x]] /;
FreeQ[{a,b},x] && IntegerQ[n] && RationalQ[m] && 0<n<=m
```

- **Reference: CRC 396, A&S 4.3.123**

```
Int[x_^m_.*Cos[a_.+b_.*x_^n_.],x_Symbol] :=
  x^(m-n+1)*Sin[a+b*x^n]/(b*n) -
  Dist[(m-n+1)/(b*n),Int[x^(m-n)*Sin[a+b*x^n],x]] /;
FreeQ[{a,b},x] && IntegerQ[n] && RationalQ[m] && 0<n<=m
```

- **Reference: CRC 405, A&S 4.3.120**
- **Derivation: Integration by parts**
- **Rule: If** $m + n + 1 = 0 \vee (n \in \mathbb{Z} \wedge ((n > 0 \wedge m < -1) \vee 0 < -n < m + 1)$**, then**

$$\int x^m \operatorname{Sin}[a + b x^n] \, dx \longrightarrow \frac{x^{m+1} \operatorname{Sin}[a + b x^n]}{m+1} - \frac{b n}{m+1} \int x^{m+n} \operatorname{Cos}[a + b x^n] \, dx$$

- **Program code:**

```
Int[x_^m_.*Sin[a_.+b_.*x_^n_.],x_Symbol] :=
  x^(m+1)*Sin[a+b*x^n]/(m+1) -
  Dist[b*n/(m+1),Int[x^(m+n)*Cos[a+b*x^n],x]] /;
FreeQ[{a,b,m,n},x] && (ZeroQ[m+n+1] || IntegerQ[n] && RationalQ[m] && (n>0 && m<-1 || 0<-n<m+1))
```

- **Reference: CRC 406, A&S 4.3.124**

```
Int[x_^m_.*Cos[a_.+b_.*x_^n_.],x_Symbol] :=
  x^(m+1)*Cos[a+b*x^n]/(m+1) +
  Dist[b*n/(m+1),Int[x^(m+n)*Sin[a+b*x^n],x]] /;
FreeQ[{a,b,m,n},x] && (ZeroQ[m+n+1] || IntegerQ[n] && RationalQ[m] && (n>0 && m<-1 || 0<-n<m+1))
```

- **Derivation: Algebraic expansion**
- **Basis:** $\mathrm{Sin}[z] = \frac{1}{2}\, i\, e^{-i z} - \frac{1}{2}\, i\, e^{i z}$
- **Rule: If** $m + 1 \neq 0 \wedge m - n + 1 \neq 0 \wedge \neg\,(m \in \mathbb{F} \vee n \in \mathbb{F} \vee n < 0)$, **then**

$$\int x^m\, \mathrm{Sin}[a + b\, x^n]\, dx \longrightarrow \frac{i}{2} \int x^m\, e^{-a\, i - b\, i\, x^n}\, dx - \frac{i}{2} \int x^m\, e^{a\, i + b\, i\, x^n}\, dx$$

- **Program code:**

```
Int[x_^m_.*Sin[a_.+b_.*x_^n_.],x_Symbol] :=
  Dist[I/2,Int[x^m*E^(-a*I-b*I*x^n),x]] -
  Dist[I/2,Int[x^m*E^(a*I+b*I*x^n),x]] /;
FreeQ[{a,b,m,n},x] && NonzeroQ[m+1] && NonzeroQ[m-n+1] &&
Not[FractionQ[m] || FractionOrNegativeQ[n]]
```

- **Basis:** $\mathrm{Cos}[z] = \frac{1}{2}\, e^{-i z} + \frac{1}{2}\, e^{i z}$

```
Int[x_^m_.*Cos[a_.+b_.*x_^n_.],x_Symbol] :=
  Dist[1/2,Int[x^m*E^(-a*I-b*I*x^n),x]] +
  Dist[1/2,Int[x^m*E^(a*I+b*I*x^n),x]] /;
FreeQ[{a,b,m,n},x] && NonzeroQ[m+1] && NonzeroQ[m-n+1] &&
Not[FractionQ[m] || FractionOrNegativeQ[n]]
```

$$\int x^m \operatorname{Sin}[a+b\,x^n]^p\,dx$$

- **Derivation: Integration by parts**
- **Rule: If $n,\ p \in \mathbb{Z} \wedge p > 1 \wedge n-1 \neq 0$, then**

$$\int \frac{\operatorname{Sin}[a+b\,x^n]^p}{x^n}\,dx \longrightarrow -\frac{\operatorname{Sin}[a+b\,x^n]^p}{(n-1)\,x^{n-1}}+\frac{b\,n\,p}{n-1}\int \operatorname{Sin}[a+b\,x^n]^{p-1}\operatorname{Cos}[a+b\,x^n]\,dx$$

- **Program code:**

```
Int[x_^m_.*Sin[a_.+b_.*x_^n_.]^p_,x_Symbol] :=
  -Sin[a+b*x^n]^p/((n-1)*x^(n-1)) +
  Dist[b*n*p/(n-1),Int[Sin[a+b*x^n]^(p-1)*Cos[a+b*x^n],x]] /;
FreeQ[{a,b},x] && IntegersQ[n,p] && ZeroQ[m+n] && p>1 && NonzeroQ[n-1]
```

```
Int[x_^m_.*Cos[a_.+b_.*x_^n_.]^p_,x_Symbol] :=
  -Cos[a+b*x^n]^p/((n-1)*x^(n-1)) -
  Dist[b*n*p/(n-1),Int[Cos[a+b*x^n]^(p-1)*Sin[a+b*x^n],x]] /;
FreeQ[{a,b},x] && IntegersQ[n,p] && ZeroQ[m+n] && p>1 && NonzeroQ[n-1]
```

- **Reference: G&R 2.631.2' special case when $m-2\,n+1=0$**
- **Rule: If $p > 1 \wedge m-2\,n+1=0$, then**

$$\int x^m \operatorname{Sin}[a+b\,x^n]^p\,dx \longrightarrow \frac{n\operatorname{Sin}[a+b\,x^n]^p}{b^2\,n^2\,p^2}-\frac{x^n\operatorname{Cos}[a+b\,x^n]\operatorname{Sin}[a+b\,x^n]^{p-1}}{b\,n\,p}+\frac{p-1}{p}\int x^m \operatorname{Sin}[a+b\,x^n]^{p-2}\,dx$$

- **Program code:**

```
Int[x_^m_.*Sin[a_.+b_.*x_^n_.]^p_,x_Symbol] :=
  n*Sin[a+b*x^n]^p/(b^2*n^2*p^2) -
  x^n*Cos[a+b*x^n]*Sin[a+b*x^n]^(p-1)/(b*n*p) +
  Dist[(p-1)/p,Int[x^m*Sin[a+b*x^n]^(p-2),x]] /;
FreeQ[{a,b,m,n},x] && RationalQ[p] && p>1 && ZeroQ[m-2*n+1]
```

- **Reference: G&R 2.631.3' special case with $m-2\,n+1=0$**

```
Int[x_^m_.*Cos[a_.+b_.*x_^n_.]^p_,x_Symbol] :=
  n*Cos[a+b*x^n]^p/(b^2*n^2*p^2) +
  x^n*Sin[a+b*x^n]*Cos[a+b*x^n]^(p-1)/(b*n*p) +
  Dist[(p-1)/p,Int[x^m*Cos[a+b*x^n]^(p-2),x]] /;
FreeQ[{a,b,m,n},x] && RationalQ[p] && p>1 && ZeroQ[m-2*n+1]
```

- **Reference: G&R 2.631.2'**
- **Rule: If** $m, n \in \mathbb{Z} \wedge p > 1 \wedge 0 < 2n < m + 1$, **then**

$$\int x^m \operatorname{Sin}[a + b x^n]^p \, dx \longrightarrow \frac{(m - n + 1) x^{m-2n+1} \operatorname{Sin}[a + b x^n]^p}{b^2 n^2 p^2} - \frac{x^{m-n+1} \operatorname{Cos}[a + b x^n] \operatorname{Sin}[a + b x^n]^{p-1}}{b n p} + \frac{p-1}{p} \int x^m \operatorname{Sin}[a + b x^n]^{p-2} \, dx - \frac{(m - n + 1)(m - 2n + 1)}{b^2 n^2 p^2} \int x^{m-2n} \operatorname{Sin}[a + b x^n]^p \, dx$$

- **Program code:**

```
Int[x_^m_.*Sin[a_.+b_.*x_^n_.]^p_,x_Symbol] :=
  (m-n+1)*x^(m-2*n+1)*Sin[a+b*x^n]^p/(b^2*n^2*p^2) -
  x^(m-n+1)*Cos[a+b*x^n]*Sin[a+b*x^n]^(p-1)/(b*n*p) +
  Dist[(p-1)/p,Int[x^m*Sin[a+b*x^n]^(p-2),x]] -
  Dist[(m-n+1)*(m-2*n+1)/(b^2*n^2*p^2),Int[x^(m-2*n)*Sin[a+b*x^n]^p,x]] /;
FreeQ[{a,b},x] && IntegersQ[m,n] && RationalQ[p] && p>1 && 0<2*n<m+1
```

- **Reference: G&R 2.631.3'**

```
Int[x_^m_.*Cos[a_.+b_.*x_^n_.]^p_,x_Symbol] :=
  (m-n+1)*x^(m-2*n+1)*Cos[a+b*x^n]^p/(b^2*n^2*p^2) +
  x^(m-n+1)*Sin[a+b*x^n]*Cos[a+b*x^n]^(p-1)/(b*n*p) +
  Dist[(p-1)/p,Int[x^m*Cos[a+b*x^n]^(p-2),x]] -
  Dist[(m-n+1)*(m-2*n+1)/(b^2*n^2*p^2),Int[x^(m-2*n)*Cos[a+b*x^n]^p,x]] /;
FreeQ[{a,b},x] && IntegersQ[m,n] && RationalQ[p] && p>1 && 0<2*n<m+1
```

- **Reference: G&R 2.643.1' special case when** $m - 2n + 1 = 0$
- **Rule: If** $p < -1 \wedge p \neq -2 \wedge m - 2n + 1 = 0$, **then**

$$\int x^m \operatorname{Sin}[a + b x^n]^p \, dx \longrightarrow \frac{x^n \operatorname{Cos}[a + b x^n] \operatorname{Sin}[a + b x^n]^{p+1}}{b n (p + 1)} - \frac{n \operatorname{Sin}[a + b x^n]^{p+2}}{b^2 n^2 (p + 1)(p + 2)} + \frac{p+2}{p+1} \int x^m \operatorname{Sin}[a + b x^n]^{p+2} \, dx$$

- **Program code:**

```
Int[x_^m_.*Sin[a_.+b_.*x_^n_.]^p_,x_Symbol] :=
  x^n*Cos[a+b*x^n]*Sin[a+b*x^n]^(p+1)/(b*n*(p+1)) -
  n*Sin[a+b*x^n]^(p+2)/(b^2*n^2*(p+1)*(p+2)) +
  Dist[(p+2)/(p+1),Int[x^m*Sin[a+b*x^n]^(p+2),x]] /;
FreeQ[{a,b,m,n},x] && RationalQ[p] && p<-1 && p≠-2 && ZeroQ[m-2*n+1]
```

- **Reference: G&R 2.643.2' special case with $m-2n+1=0$**

```
Int[x_^m_.*Cos[a_.+b_.*x_^n_.]^p_,x_Symbol] :=
  -x^n*Sin[a+b*x^n]*Cos[a+b*x^n]^(p+1)/(b*n*(p+1)) -
  n*Cos[a+b*x^n]^(p+2)/(b^2*n^2*(p+1)*(p+2)) +
  Dist[(p+2)/(p+1),Int[x^m*Cos[a+b*x^n]^(p+2),x]] /;
FreeQ[{a,b,m,n},x] && RationalQ[p] && p<-1 && p≠-2 && ZeroQ[m-2*n+1]
```

- **Reference: G&R 2.643.1'**

- **Rule: If $m, n \in \mathbb{Z} \wedge p < -1 \wedge p \neq -2 \wedge 0 < 2n < m+1$, then**

$$\int x^m \operatorname{Sin}[a+b\,x^n]^p\,dx \longrightarrow \frac{x^{m-n+1}\operatorname{Cos}[a+b\,x^n]\operatorname{Sin}[a+b\,x^n]^{p+1}}{b\,n\,(p+1)} - \frac{(m-n+1)\,x^{m-2n+1}\operatorname{Sin}[a+b\,x^n]^{p+2}}{b^2\,n^2\,(p+1)\,(p+2)} + \frac{p+2}{p+1}\int x^m \operatorname{Sin}[a+b\,x^n]^{p+2}\,dx + \frac{(m-n+1)\,(m-2n+1)}{b^2\,n^2\,(p+1)\,(p+2)}\int x^{m-2n}\operatorname{Sin}[a+b\,x^n]^{p+2}\,dx$$

- **Program code:**

```
Int[x_^m_.*Sin[a_.+b_.*x_^n_.]^p_,x_Symbol] :=
  x^(m-n+1)*Cos[a+b*x^n]*Sin[a+b*x^n]^(p+1)/(b*n*(p+1)) -
  (m-n+1)*x^(m-2*n+1)*Sin[a+b*x^n]^(p+2)/(b^2*n^2*(p+1)*(p+2)) +
  Dist[(p+2)/(p+1),Int[x^m*Sin[a+b*x^n]^(p+2),x]] +
  Dist[(m-n+1)*(m-2*n+1)/(b^2*n^2*(p+1)*(p+2)),Int[x^(m-2*n)*Sin[a+b*x^n]^(p+2),x]] /;
FreeQ[{a,b},x] && IntegersQ[m,n] && RationalQ[p] && p<-1 && p≠-2 && 0<2*n<m+1
```

- **Reference: G&R 2.643.2**

```
Int[x_^m_.*Cos[a_.+b_.*x_^n_.]^p_,x_Symbol] :=
  -x^(m-n+1)*Sin[a+b*x^n]*Cos[a+b*x^n]^(p+1)/(b*n*(p+1)) -
  (m-n+1)*x^(m-2*n+1)*Cos[a+b*x^n]^(p+2)/(b^2*n^2*(p+1)*(p+2)) +
  Dist[(p+2)/(p+1),Int[x^m*Cos[a+b*x^n]^(p+2),x]] +
  Dist[(m-n+1)*(m-2*n+1)/(b^2*n^2*(p+1)*(p+2)),Int[x^(m-2*n)*Cos[a+b*x^n]^(p+2),x]] /;
FreeQ[{a,b},x] && IntegersQ[m,n] && RationalQ[p] && p<-1 && p≠-2 && 0<2*n<m+1
```

- **Reference: G&R 2.638.1'**

- **Rule: If $m, n \in \mathbb{Z} \land p > 1 \land 0 < 2n < 1 - m \land m + n + 1 \neq 0$, then**

$$\int x^m \operatorname{Sin}[a+b\,x^n]^p\,dx \longrightarrow \frac{x^{m+1}\operatorname{Sin}[a+b\,x^n]^p}{m+1} - \frac{b\,n\,p\,x^{m+n+1}\operatorname{Cos}[a+b\,x^n]\operatorname{Sin}[a+b\,x^n]^{p-1}}{(m+1)\,(m+n+1)} - \\ \frac{b^2\,n^2\,p^2}{(m+1)\,(m+n+1)}\int x^{m+2n}\operatorname{Sin}[a+b\,x^n]^p\,dx + \frac{b^2\,n^2\,p\,(p-1)}{(m+1)\,(m+n+1)}\int x^{m+2n}\operatorname{Sin}[a+b\,x^n]^{p-2}\,dx$$

- **Program code:**

```
Int[x_^m_.*Sin[a_.+b_.*x_^n_.]^p_,x_Symbol] :=
  x^(m+1)*Sin[a+b*x^n]^p/(m+1) -
  b*n*p*x^(m+n+1)*Cos[a+b*x^n]*Sin[a+b*x^n]^(p-1)/((m+1)*(m+n+1)) -
  Dist[b^2*n^2*p^2/((m+1)*(m+n+1)),Int[x^(m+2*n)*Sin[a+b*x^n]^p,x]] +
  Dist[b^2*n^2*p*(p-1)/((m+1)*(m+n+1)),Int[x^(m+2*n)*Sin[a+b*x^n]^(p-2),x]] /;
FreeQ[{a,b},x] && IntegersQ[m,n] && RationalQ[p] && p>1 && 0<2*n<1-m && NonzeroQ[m+n+1]
```

- **Reference: G&R 2.638.2'**

```
Int[x_^m_.*Cos[a_.+b_.*x_^n_.]^p_,x_Symbol] :=
  x^(m+1)*Cos[a+b*x^n]^p/(m+1) +
  b*n*p*x^(m+n+1)*Sin[a+b*x^n]*Cos[a+b*x^n]^(p-1)/((m+1)*(m+n+1)) -
  Dist[b^2*n^2*p^2/((m+1)*(m+n+1)),Int[x^(m+2*n)*Cos[a+b*x^n]^p,x]] +
  Dist[b^2*n^2*p*(p-1)/((m+1)*(m+n+1)),Int[x^(m+2*n)*Cos[a+b*x^n]^(p-2),x]] /;
FreeQ[{a,b},x] && IntegersQ[m,n] && RationalQ[p] && p>1 && 0<2*n<1-m && NonzeroQ[m+n+1]
```

- **Derivation: Algebraic expansion**

- **Basis: $\operatorname{Sin}[z] = \frac{1}{2}\,i\,e^{-i\,z} - \frac{1}{2}\,i\,e^{i\,z}$**

- **Note: Not sure if this is useful or necessary.**

- **Rule: If $p \in \mathbb{Z} \land p > 0 \land m + 1 \neq 0 \land m - n + 1 \neq 0$, then**

$$\int x^m \operatorname{Sin}[a+b\,x^n]^p\,dx \longrightarrow \left(\frac{i}{2}\right)^p \int x^m \left(e^{-a\,i-b\,i\,x^n} - e^{a\,i+b\,i\,x^n}\right)^p dx$$

- **Program code:**

```
(* Int[x_^m_.*Sin[a_.+b_.*x_^n_.]^p_,x_Symbol] :=
  Dist[(I/2)^p,Int[x^m*(E^(-a*I-b*I*x^n)-E^(a*I+b*I*x^n))^p,x]] /;
FreeQ[{a,b,m,n},x] && IntegerQ[p] && p>0 && NonzeroQ[m+1] && NonzeroQ[m-n+1] && Not[FractionQ[m] || 
```

$$\int x^m \operatorname{Sin}[a+b\,(c+d\,x)^n]^p\,dx$$

- **Derivation: Integration by linear substitution**
- **Rule: If** $m\in\mathbb{Z}\bigwedge m>0\bigwedge p\in\mathbb{Q}$**, then**

$$\int x^m \operatorname{Sin}[a+b\,(c+d\,x)^n]^p\,dx \longrightarrow \frac{1}{d}\operatorname{Subst}\left[\int\left(-\frac{c}{d}+\frac{x}{d}\right)^m \operatorname{Sin}[a+b\,x^n]^p\,dx,\ x,\ c+d\,x\right]$$

- **Program code:**

```
Int[x_^m_.*Sin[a_.+b_.*(c_+d_.*x_)^n_]^p_.,x_Symbol] :=
  Dist[1/d,Subst[Int[(-c/d+x/d)^m*Sin[a+b*x^n]^p,x],x,c+d*x]] /;
FreeQ[{a,b,c,d,n},x] && IntegerQ[m] && m>0 && RationalQ[p]
```

```
Int[x_^m_.*Cos[a_.+b_.*(c_+d_.*x_)^n_]^p_.,x_Symbol] :=
  Dist[1/d,Subst[Int[(-c/d+x/d)^m*Cos[a+b*x^n]^p,x],x,c+d*x]] /;
FreeQ[{a,b,c,d,n},x] && IntegerQ[m] && m>0 && RationalQ[p]
```

$$\int \mathrm{Sin}\left[a+b\,x+c\,x^2\right]\,dx$$

- **Derivation: Algebraic simplification**
- **Basis: If** $b^2-4\,a\,c=0$**, then** $a+b\,x+c\,x^2=\frac{(b+2\,c\,x)^2}{4\,c}$
- **Rule: If** $b^2-4\,a\,c=0$**, then**

$$\int \mathrm{Sin}\left[a+b\,x+c\,x^2\right]\,dx \longrightarrow \int \mathrm{Sin}\left[\frac{(b+2\,c\,x)^2}{4\,c}\right]\,dx$$

- **Program code:**

```
Int[Sin[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  Int[Sin[(b+2*c*x)^2/(4*c)],x] /;
FreeQ[{a,b,c},x] && ZeroQ[b^2-4*a*c]
```

```
Int[Cos[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  Int[Cos[(b+2*c*x)^2/(4*c)],x] /;
FreeQ[{a,b,c},x] && ZeroQ[b^2-4*a*c]
```

- **Derivation: Algebraic expansion**
- **Basis:** $a+b\,x+c\,x^2=\frac{(b+2\,c\,x)^2}{4\,c}-\frac{b^2-4\,a\,c}{4\,c}$
- **Basis:** $\mathrm{Sin}[z-w]=\mathrm{Cos}[w]\,\mathrm{Sin}[z]-\mathrm{Sin}[w]\,\mathrm{Cos}[z]$
- **Rule: If** $b^2-4\,a\,c\neq 0$**, then**

$$\int \mathrm{Sin}\left[a+b\,x+c\,x^2\right]\,dx \longrightarrow \mathrm{Cos}\left[\frac{b^2-4\,a\,c}{4\,c}\right]\int \mathrm{Sin}\left[\frac{(b+2\,c\,x)^2}{4\,c}\right]\,dx-\mathrm{Sin}\left[\frac{b^2-4\,a\,c}{4\,c}\right]\int \mathrm{Cos}\left[\frac{(b+2\,c\,x)^2}{4\,c}\right]\,dx$$

- **Program code:**

```
Int[Sin[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  Cos[(b^2-4*a*c)/(4*c)]*Int[Sin[(b+2*c*x)^2/(4*c)],x] -
  Sin[(b^2-4*a*c)/(4*c)]*Int[Cos[(b+2*c*x)^2/(4*c)],x] /;
FreeQ[{a,b,c},x] && NonzeroQ[b^2-4*a*c]
```

```
Int[Cos[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  Cos[(b^2-4*a*c)/(4*c)]*Int[Cos[(b+2*c*x)^2/(4*c)],x] +
  Sin[(b^2-4*a*c)/(4*c)]*Int[Sin[(b+2*c*x)^2/(4*c)],x] /;
FreeQ[{a,b,c},x] && NonzeroQ[b^2-4*a*c]
```

$$\int (d+e\,x)^m \operatorname{Sin}\left[a+b\,x+c\,x^2\right] dx$$

- **Rule: If** $b\,e-2\,c\,d=0$**, then**

$$\int (d+e\,x) \operatorname{Sin}\left[a+b\,x+c\,x^2\right] dx \longrightarrow -\frac{e\operatorname{Cos}\left[a+b\,x+c\,x^2\right]}{2\,c}$$

- **Program code:**

```
Int[(d_.+e_.*x_)*Sin[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  -e*Cos[a+b*x+c*x^2]/(2*c) /;
FreeQ[{a,b,c,d,e},x] && ZeroQ[b*e-2*c*d]
```

```
Int[(d_.+e_.*x_)*Cos[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  e*Sin[a+b*x+c*x^2]/(2*c) /;
FreeQ[{a,b,c,d,e},x] && ZeroQ[b*e-2*c*d]
```

- **Rule: If** $b\,e-2\,c\,d\neq 0$**, then**

$$\int (d+e\,x) \operatorname{Sin}\left[a+b\,x+c\,x^2\right] dx \longrightarrow -\frac{e\operatorname{Cos}\left[a+b\,x+c\,x^2\right]}{2\,c}-\frac{b\,e-2\,c\,d}{2\,c}\int \operatorname{Sin}\left[a+b\,x+c\,x^2\right] dx$$

- **Program code:**

```
Int[(d_.+e_.*x_)*Sin[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  -e*Cos[a+b*x+c*x^2]/(2*c) -
  Dist[(b*e-2*c*d)/(2*c),Int[Sin[a+b*x+c*x^2],x]] /;
FreeQ[{a,b,c,d,e},x] && NonzeroQ[b*e-2*c*d]
```

```
Int[(d_.+e_.*x_)*Cos[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  e*Sin[a+b*x+c*x^2]/(2*c) -
  Dist[(b*e-2*c*d)/(2*c),Int[Cos[a+b*x+c*x^2],x]] /;
FreeQ[{a,b,c,d,e},x] && NonzeroQ[b*e-2*c*d]
```

- **Rule: If $m > 1 \land b e - 2 c d = 0$, then**

$$\int (d + e x)^m \operatorname{Sin}\left[a + b x + c x^2\right] dx \longrightarrow -\frac{e (d + e x)^{m-1} \operatorname{Cos}\left[a + b x + c x^2\right]}{2 c} + \frac{e^2 (m - 1)}{2 c} \int (d + e x)^{m-2} \operatorname{Cos}\left[a + b x + c x^2\right] dx$$

- **Program code:**

```
Int[(d_.+e_.*x_)^m_*Sin[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  -e*(d+e*x)^(m-1)*Cos[a+b*x+c*x^2]/(2*c) +
  Dist[e^2*(m-1)/(2*c),Int[(d+e*x)^(m-2)*Cos[a+b*x+c*x^2],x]] /;
FreeQ[{a,b,c,d,e},x] && RationalQ[m] && m>1 && ZeroQ[b*e-2*c*d]
```

```
Int[(d_.+e_.*x_)^m_*Cos[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  e*(d+e*x)^(m-1)*Sin[a+b*x+c*x^2]/(2*c) -
  Dist[e^2*(m-1)/(2*c),Int[(d+e*x)^(m-2)*Sin[a+b*x+c*x^2],x]] /;
FreeQ[{a,b,c,d,e},x] && RationalQ[m] && m>1 && ZeroQ[b*e-2*c*d]
```

- **Rule: If $m > 1 \land b e - 2 c d \neq 0$, then**

$$\int (d + e x)^m \operatorname{Sin}\left[a + b x + c x^2\right] dx \longrightarrow -\frac{e (d + e x)^{m-1} \operatorname{Cos}\left[a + b x + c x^2\right]}{2 c} - \frac{b e - 2 c d}{2 c} \int (d + e x)^{m-1} \operatorname{Sin}\left[a + b x + c x^2\right] dx + \frac{e^2 (m - 1)}{2 c} \int (d + e x)^{m-2} \operatorname{Cos}\left[a + b x + c x^2\right] dx$$

- **Program code:**

```
Int[(d_.+e_.*x_)^m_*Sin[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  -e*(d+e*x)^(m-1)*Cos[a+b*x+c*x^2]/(2*c) -
  Dist[(b*e-2*c*d)/(2*c),Int[(d+e*x)^(m-1)*Sin[a+b*x+c*x^2],x]] +
  Dist[e^2*(m-1)/(2*c),Int[(d+e*x)^(m-2)*Cos[a+b*x+c*x^2],x]] /;
FreeQ[{a,b,c,d,e},x] && RationalQ[m] && m>1 && NonzeroQ[b*e-2*c*d]
```

```
Int[(d_.+e_.*x_)^m_*Cos[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  e*(d+e*x)^(m-1)*Sin[a+b*x+c*x^2]/(2*c) -
  Dist[(b*e-2*c*d)/(2*c),Int[(d+e*x)^(m-1)*Cos[a+b*x+c*x^2],x]] -
  Dist[e^2*(m-1)/(2*c),Int[(d+e*x)^(m-2)*Sin[a+b*x+c*x^2],x]] /;
FreeQ[{a,b,c,d,e},x] && RationalQ[m] && m>1 && NonzeroQ[b*e-2*c*d]
```

- **Rule: If** $m < -1 \wedge b e - 2 c d = 0$**, then**

$$\int (d+e x)^m \operatorname{Sin}\left[a+b x+c x^2\right] dx \longrightarrow \frac{(d+e x)^{m+1} \operatorname{Sin}\left[a+b x+c x^2\right]}{e (m+1)} - \frac{2 c}{e^2 (m+1)} \int (d+e x)^{m+2} \operatorname{Cos}\left[a+b x+c x^2\right] dx$$

- **Program code:**

```
Int[(d_.+e_.*x_)^m_*Sin[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  (d+e*x)^(m+1)*Sin[a+b*x+c*x^2]/(e*(m+1)) -
  Dist[2*c/(e^2*(m+1)),Int[(d+e*x)^(m+2)*Cos[a+b*x+c*x^2],x]] /;
FreeQ[{a,b,c,d,e},x] && RationalQ[m] && m<-1 && ZeroQ[b*e-2*c*d]
```

```
Int[(d_.+e_.*x_)^m_*Cos[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  (d+e*x)^(m+1)*Cos[a+b*x+c*x^2]/(e*(m+1)) +
  Dist[2*c/(e^2*(m+1)),Int[(d+e*x)^(m+2)*Sin[a+b*x+c*x^2],x]] /;
FreeQ[{a,b,c,d,e},x] && RationalQ[m] && m<-1 && ZeroQ[b*e-2*c*d]
```

- **Rule: If** $m < -1 \wedge b e - 2 c d \neq 0$**, then**

$$\int (d+e x)^m \operatorname{Sin}\left[a+b x+c x^2\right] dx \longrightarrow \frac{(d+e x)^{m+1} \operatorname{Sin}\left[a+b x+c x^2\right]}{e (m+1)} - \frac{b e - 2 c d}{e^2 (m+1)} \int (d+e x)^{m+1} \operatorname{Cos}\left[a+b x+c x^2\right] dx - \frac{2 c}{e^2 (m+1)} \int (d+e x)^{m+2} \operatorname{Cos}\left[a+b x+c x^2\right] dx$$

- **Program code:**

```
Int[(d_.+e_.*x_)^m_*Sin[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  (d+e*x)^(m+1)*Sin[a+b*x+c*x^2]/(e*(m+1)) -
  Dist[(b*e-2*c*d)/(e^2*(m+1)),Int[(d+e*x)^(m+1)*Cos[a+b*x+c*x^2],x]] -
  Dist[2*c/(e^2*(m+1)),Int[(d+e*x)^(m+2)*Cos[a+b*x+c*x^2],x]] /;
FreeQ[{a,b,c,d,e},x] && RationalQ[m] && m<-1 && NonzeroQ[b*e-2*c*d]
```

```
Int[(d_.+e_.*x_)^m_*Cos[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  (d+e*x)^(m+1)*Cos[a+b*x+c*x^2]/(e*(m+1)) +
  Dist[(b*e-2*c*d)/(e^2*(m+1)),Int[(d+e*x)^(m+1)*Sin[a+b*x+c*x^2],x]] +
  Dist[2*c/(e^2*(m+1)),Int[(d+e*x)^(m+2)*Sin[a+b*x+c*x^2],x]] /;
FreeQ[{a,b,c,d,e},x] && RationalQ[m] && m<-1 && NonzeroQ[b*e-2*c*d]
```

$$\int \mathrm{Sin}[a+b\,\mathrm{Log}[c\,x^n]]^p\,dx$$

- **Rule: If** $1+b^2 n^2 \neq 0$, **then**

$$\int \mathrm{Sin}[a+b\,\mathrm{Log}[c\,x^n]]\,dx \longrightarrow \frac{x\,\mathrm{Sin}[a+b\,\mathrm{Log}[c\,x^n]]}{1+b^2 n^2}-\frac{b\,n\,x\,\mathrm{Cos}[a+b\,\mathrm{Log}[c\,x^n]]}{1+b^2 n^2}$$

- **Program code:**

```
Int[Sin[a_.+b_.*Log[c_.*x_^n_.]],x_Symbol] :=
  x*Sin[a+b*Log[c*x^n]]/(1+b^2*n^2) -
  b*n*x*Cos[a+b*Log[c*x^n]]/(1+b^2*n^2) /;
FreeQ[{a,b,c,n},x] && NonzeroQ[1+b^2*n^2]
```

```
Int[Cos[a_.+b_.*Log[c_.*x_^n_.]],x_Symbol] :=
  x*Cos[a+b*Log[c*x^n]]/(1+b^2*n^2) +
  b*n*x*Sin[a+b*Log[c*x^n]]/(1+b^2*n^2) /;
FreeQ[{a,b,c,n},x] && NonzeroQ[1+b^2*n^2]
```

- **Rule: If** $p>1 \wedge 1+b^2 n^2 p^2 \neq 0$, **then**

$$\int \mathrm{Sin}[a+b\,\mathrm{Log}[c\,x^n]]^p\,dx \longrightarrow \frac{x\,\mathrm{Sin}[a+b\,\mathrm{Log}[c\,x^n]]^p}{1+b^2 n^2 p^2}-$$
$$\frac{b\,n\,p\,x\,\mathrm{Cos}[a+b\,\mathrm{Log}[c\,x^n]]\,\mathrm{Sin}[a+b\,\mathrm{Log}[c\,x^n]]^{p-1}}{1+b^2 n^2 p^2}+\frac{b^2 n^2 p\,(p-1)}{1+b^2 n^2 p^2}\int \mathrm{Sin}[a+b\,\mathrm{Log}[c\,x^n]]^{p-2}\,dx$$

- **Program code:**

```
Int[Sin[a_.+b_.*Log[c_.*x_^n_.]]^p_,x_Symbol] :=
  x*Sin[a+b*Log[c*x^n]]^p/(1+b^2*n^2*p^2) -
  b*n*p*x*Cos[a+b*Log[c*x^n]]*Sin[a+b*Log[c*x^n]]^(p-1)/(1+b^2*n^2*p^2) +
  Dist[b^2*n^2*p*(p-1)/(1+b^2*n^2*p^2),Int[Sin[a+b*Log[c*x^n]]^(p-2),x]] /;
FreeQ[{a,b,c,n},x] && RationalQ[p] && p>1 && NonzeroQ[1+b^2*n^2*p^2]
```

```
Int[Cos[a_.+b_.*Log[c_.*x_^n_.]]^p_,x_Symbol] :=
  x*Cos[a+b*Log[c*x^n]]^p/(1+b^2*n^2*p^2) +
  b*n*p*x*Cos[a+b*Log[c*x^n]]^(p-1)*Sin[a+b*Log[c*x^n]]/(1+b^2*n^2*p^2) +
  Dist[b^2*n^2*p*(p-1)/(1+b^2*n^2*p^2),Int[Cos[a+b*Log[c*x^n]]^(p-2),x]] /;
FreeQ[{a,b,c,n},x] && RationalQ[p] && p>1 && NonzeroQ[1+b^2*n^2*p^2]
```

- **Rule: If** $p \neq -1 \land p \neq -2 \land 1 + b^2 n^2 (p+2)^2 = 0$**, then**

$$\int \text{Sin}[a + b\,\text{Log}[c\,x^n]]^p\,dx \longrightarrow \frac{x\,\text{Cot}[a + b\,\text{Log}[c\,x^n]]\,\text{Sin}[a + b\,\text{Log}[c\,x^n]]^{p+2}}{b\,n\,(p+1)} - \frac{x\,\text{Sin}[a + b\,\text{Log}[c\,x^n]]^{p+2}}{b^2\,n^2\,(p+1)\,(p+2)}$$

- **Program code:**

```
Int[Sin[a_.+b_.*Log[c_.*x_^n_.]]^p_,x_Symbol] :=
  x*Cot[a+b*Log[c*x^n]]*Sin[a+b*Log[c*x^n]]^(p+2)/(b*n*(p+1)) -
  x*Sin[a+b*Log[c*x^n]]^(p+2)/(b^2*n^2*(p+1)*(p+2)) /;
FreeQ[{a,b,c,n,p},x] && NonzeroQ[p+1] && NonzeroQ[p+2] && ZeroQ[1+b^2*n^2*(p+2)^2]
```

```
Int[Cos[a_.+b_.*Log[c_.*x_^n_.]]^p_,x_Symbol] :=
  -x*Tan[a+b*Log[c*x^n]]*Cos[a+b*Log[c*x^n]]^(p+2)/(b*n*(p+1)) -
  x*Cos[a+b*Log[c*x^n]]^(p+2)/(b^2*n^2*(p+1)*(p+2)) /;
FreeQ[{a,b,c,n,p},x] && NonzeroQ[p+1] && NonzeroQ[p+2] && ZeroQ[1+b^2*n^2*(p+2)^2]
```

- **Rule: If** $p < -1 \land p \neq -2 \land 1 + b^2 n^2 (p+2)^2 \neq 0$**, then**

$$\int \text{Sin}[a + b\,\text{Log}[c\,x^n]]^p\,dx \longrightarrow \frac{x\,\text{Cot}[a + b\,\text{Log}[c\,x^n]]\,\text{Sin}[a + b\,\text{Log}[c\,x^n]]^{p+2}}{b\,n\,(p+1)} - \frac{x\,\text{Sin}[a + b\,\text{Log}[c\,x^n]]^{p+2}}{b^2\,n^2\,(p+1)\,(p+2)} + \frac{1 + b^2\,n^2\,(p+2)^2}{b^2\,n^2\,(p+1)\,(p+2)} \int \text{Sin}[a + b\,\text{Log}[c\,x^n]]^{p+2}\,dx$$

- **Program code:**

```
Int[Sin[a_.+b_.*Log[c_.*x_^n_.]]^p_,x_Symbol] :=
  x*Cot[a+b*Log[c*x^n]]*Sin[a+b*Log[c*x^n]]^(p+2)/(b*n*(p+1)) -
  x*Sin[a+b*Log[c*x^n]]^(p+2)/(b^2*n^2*(p+1)*(p+2)) +
  Dist[(1+b^2*n^2*(p+2)^2)/(b^2*n^2*(p+1)*(p+2)),Int[Sin[a+b*Log[c*x^n]]^(p+2),x]] /;
FreeQ[{a,b,c,n},x] && RationalQ[p] && p<-1 && p≠-2 && NonzeroQ[1+b^2*n^2*(p+2)^2]
```

```
Int[Cos[a_.+b_.*Log[c_.*x_^n_.]]^p_,x_Symbol] :=
  -x*Tan[a+b*Log[c*x^n]]*Cos[a+b*Log[c*x^n]]^(p+2)/(b*n*(p+1)) -
  x*Cos[a+b*Log[c*x^n]]^(p+2)/(b^2*n^2*(p+1)*(p+2)) +
  Dist[(1+b^2*n^2*(p+2)^2)/(b^2*n^2*(p+1)*(p+2)),Int[Cos[a+b*Log[c*x^n]]^(p+2),x]] /;
FreeQ[{a,b,c,n},x] && RationalQ[p] && p<-1 && p≠-2 && NonzeroQ[1+b^2*n^2*(p+2)^2]
```

$$\int x^m \operatorname{Sin}[a + b \operatorname{Log}[c\, x^n]]^p \, dx$$

- **Rule: If** $b^2 n^2 + (m+1)^2 \neq 0 \land m+1 \neq 0$, **then**

$$\int x^m \operatorname{Sin}[a + b \operatorname{Log}[c\, x^n]] \, dx \longrightarrow \frac{(m+1)\, x^{m+1} \operatorname{Sin}[a + b \operatorname{Log}[c\, x^n]]}{b^2 n^2 + (m+1)^2} - \frac{b\, n\, x^{m+1} \operatorname{Cos}[a + b \operatorname{Log}[c\, x^n]]}{b^2 n^2 + (m+1)^2}$$

- **Program code:**

```
Int[x_^m_.*Sin[a_.+b_.*Log[c_.*x_^n_.]],x_Symbol] :=
  (m+1)*x^(m+1)*Sin[a+b*Log[c*x^n]]/(b^2*n^2+(m+1)^2) -
  b*n*x^(m+1)*Cos[a+b*Log[c*x^n]]/(b^2*n^2+(m+1)^2) /;
FreeQ[{a,b,c,m,n},x] && NonzeroQ[b^2*n^2+(m+1)^2] && NonzeroQ[m+1]
```

```
Int[x_^m_.*Cos[a_.+b_.*Log[c_.*x_^n_.]],x_Symbol] :=
  (m+1)*x^(m+1)*Cos[a+b*Log[c*x^n]]/(b^2*n^2+(m+1)^2) +
  b*n*x^(m+1)*Sin[a+b*Log[c*x^n]]/(b^2*n^2+(m+1)^2) /;
FreeQ[{a,b,c,m,n},x] && NonzeroQ[b^2*n^2+(m+1)^2] && NonzeroQ[m+1]
```

- **Rule: If** $p > 1 \land b^2 n^2 p^2 + (m+1)^2 \neq 0 \land m+1 \neq 0$, **then**

$$\int x^m \operatorname{Sin}[a + b \operatorname{Log}[c\, x^n]]^p \, dx \longrightarrow \frac{(m+1)\, x^{m+1} \operatorname{Sin}[a + b \operatorname{Log}[c\, x^n]]^p}{b^2 n^2 p^2 + (m+1)^2} -$$
$$\frac{b\, n\, p\, x^{m+1} \operatorname{Cos}[a + b \operatorname{Log}[c\, x^n]] \operatorname{Sin}[a + b \operatorname{Log}[c\, x^n]]^{p-1}}{b^2 n^2 p^2 + (m+1)^2} + \frac{b^2 n^2 p\, (p-1)}{b^2 n^2 p^2 + (m+1)^2} \int x^m \operatorname{Sin}[a + b \operatorname{Log}[c\, x^n]]^{p-2} \, dx$$

- **Program code:**

```
Int[x_^m_.*Sin[a_.+b_.*Log[c_.*x_^n_.]]^p_,x_Symbol] :=
  (m+1)*x^(m+1)*Sin[a+b*Log[c*x^n]]^p/(b^2*n^2*p^2+(m+1)^2) -
  b*n*p*x^(m+1)*Cos[a+b*Log[c*x^n]]*Sin[a+b*Log[c*x^n]]^(p-1)/(b^2*n^2*p^2+(m+1)^2) +
  Dist[b^2*n^2*p*(p-1)/(b^2*n^2*p^2+(m+1)^2),Int[x^m*Sin[a+b*Log[c*x^n]]^(p-2),x]] /;
FreeQ[{a,b,c,m,n},x] && RationalQ[p] && p>1 && NonzeroQ[b^2*n^2*p^2+(m+1)^2] && NonzeroQ[m+1]
```

```
Int[x_^m_.*Cos[a_.+b_.*Log[c_.*x_^n_.]]^p_,x_Symbol] :=
  (m+1)*x^(m+1)*Cos[a+b*Log[c*x^n]]^p/(b^2*n^2*p^2+(m+1)^2) +
  b*n*p*x^(m+1)*Sin[a+b*Log[c*x^n]]*Cos[a+b*Log[c*x^n]]^(p-1)/(b^2*n^2*p^2+(m+1)^2) +
  Dist[b^2*n^2*p*(p-1)/(b^2*n^2*p^2+(m+1)^2),Int[x^m*Cos[a+b*Log[c*x^n]]^(p-2),x]] /;
FreeQ[{a,b,c,m,n},x] && RationalQ[p] && p>1 && NonzeroQ[b^2*n^2*p^2+(m+1)^2] && NonzeroQ[m+1]
```

- **Rule: If** $p < -1 \land p \neq -2 \land m + 1 \neq 0$, **then**

$$\int x^m \operatorname{Sin}[a + b \operatorname{Log}[c\, x^n]]^p \, dx \longrightarrow \frac{x^{m+1} \operatorname{Cot}[a + b \operatorname{Log}[c\, x^n]] \operatorname{Sin}[a + b \operatorname{Log}[c\, x^n]]^{p+2}}{b\, n\, (p + 1)} -$$

$$\frac{(m + 1)\, x^{m+1} \operatorname{Sin}[a + b \operatorname{Log}[c\, x^n]]^{p+2}}{b^2\, n^2\, (p + 1)\, (p + 2)} + \frac{b^2\, n^2\, (p + 2)^2 + (m + 1)^2}{b^2\, n^2\, (p + 1)\, (p + 2)} \int x^m \operatorname{Sin}[a + b \operatorname{Log}[c\, x^n]]^{p+2} \, dx$$

- **Program code:**

```
Int[x_^m_.*Sin[a_.+b_.*Log[c_.*x_^n_.]]^p_,x_Symbol] :=
  x^(m+1)*Cot[a+b*Log[c*x^n]]*Sin[a+b*Log[c*x^n]]^(p+2)/(b*n*(p+1)) -
  (m+1)*x^(m+1)*Sin[a+b*Log[c*x^n]]^(p+2)/(b^2*n^2*(p+1)*(p+2)) +
  Dist[(b^2*n^2*(p+2)^2+(m+1)^2)/(b^2*n^2*(p+1)*(p+2)),Int[x^m*Sin[a+b*Log[c*x^n]]^(p+2),x]] /;
FreeQ[{a,b,c,m,n},x] && RationalQ[p] && p<-1 && p≠-2 && NonzeroQ[m+1]
```

```
Int[x_^m_.*Cos[a_.+b_.*Log[c_.*x_^n_.]]^p_,x_Symbol] :=
  -x^(m+1)*Tan[a+b*Log[c*x^n]]*Cos[a+b*Log[c*x^n]]^(p+2)/(b*n*(p+1)) -
  (m+1)*x^(m+1)*Cos[a+b*Log[c*x^n]]^(p+2)/(b^2*n^2*(p+1)*(p+2)) +
  Dist[(b^2*n^2*(p+2)^2+(m+1)^2)/(b^2*n^2*(p+1)*(p+2)),Int[x^m*Cos[a+b*Log[c*x^n]]^(p+2),x]] /;
FreeQ[{a,b,c,m,n},x] && RationalQ[p] && p<-1 && p≠-2 && NonzeroQ[m+1]
```

$$\int \text{Sin}[a\, x^n\, \text{Log}[b\, x]^p]\, \text{Log}[b\, x]^p\, dx$$

- **Rule: If $p > 0$, then**

$$\int \text{Sin}[a\, x\, \text{Log}[b\, x]^p]\, \text{Log}[b\, x]^p\, dx \longrightarrow -\frac{\text{Cos}[a\, x\, \text{Log}[b\, x]^p]}{a} - p \int \text{Sin}[a\, x\, \text{Log}[b\, x]^p]\, \text{Log}[b\, x]^{p-1}\, dx$$

- **Program code:**

```
Int[Sin[a_.*x_*Log[b_.*x_]^p_.]*Log[b_.*x_]^p_.,x_Symbol] :=
  -Cos[a*x*Log[b*x]^p]/a -
  Dist[p,Int[Sin[a*x*Log[b*x]^p]*Log[b*x]^(p-1),x]] /;
FreeQ[{a,b},x] && RationalQ[p] && p>0
```

```
Int[Cos[a_.*x_*Log[b_.*x_]^p_.]*Log[b_.*x_]^p_.,x_Symbol] :=
  Sin[a*x*Log[b*x]^p]/a -
  Dist[p,Int[Cos[a*x*Log[b*x]^p]*Log[b*x]^(p-1),x]] /;
FreeQ[{a,b},x] && RationalQ[p] && p>0
```

- **Rule: If $p > 0$, then**

$$\int \text{Sin}[a\, x^n\, \text{Log}[b\, x]^p]\, \text{Log}[b\, x]^p\, dx \longrightarrow -\frac{\text{Cos}[a\, x^n\, \text{Log}[b\, x]^p]}{a\, n\, x^{n-1}} - \frac{p}{n} \int \text{Sin}[a\, x^n\, \text{Log}[b\, x]^p]\, \text{Log}[b\, x]^{p-1}\, dx - \frac{n-1}{a\, n} \int \frac{\text{Cos}[a\, x^n\, \text{Log}[b\, x]^p]}{x^n}\, dx$$

- **Program code:**

```
Int[Sin[a_.*x_^n_*Log[b_.*x_]^p_.]*Log[b_.*x_]^p_.,x_Symbol] :=
  -Cos[a*x^n*Log[b*x]^p]/(a*n*x^(n-1)) -
  Dist[p/n,Int[Sin[a*x^n*Log[b*x]^p]*Log[b*x]^(p-1),x]] -
  Dist[(n-1)/(a*n),Int[Cos[a*x^n*Log[b*x]^p]/x^n,x]] /;
FreeQ[{a,b},x] && RationalQ[{n,p}] && p>0
```

```
Int[Cos[a_.*x_^n_*Log[b_.*x_]^p_.]*Log[b_.*x_]^p_.,x_Symbol] :=
  Sin[a*x^n*Log[b*x]^p]/(a*n*x^(n-1)) -
  Dist[p/n,Int[Cos[a*x^n*Log[b*x]^p]*Log[b*x]^(p-1),x]] +
  Dist[(n-1)/(a*n),Int[Sin[a*x^n*Log[b*x]^p]/x^n,x]] /;
FreeQ[{a,b},x] && RationalQ[{n,p}] && p>0
```

$$\int x^m \operatorname{Sin}[a\,x^n \operatorname{Log}[b\,x]^p]\,\operatorname{Log}[b\,x]^p\,dx$$

- **Rule: If $p > 0 \land m - n + 1 = 0$, then**

$$\int x^m \operatorname{Sin}[a\,x^n \operatorname{Log}[b\,x]^p]\,\operatorname{Log}[b\,x]^p\,dx \longrightarrow -\frac{\operatorname{Cos}[a\,x^n \operatorname{Log}[b\,x]^p]}{a\,n} - \frac{p}{n}\int x^m \operatorname{Sin}[a\,x^n \operatorname{Log}[b\,x]^p]\,\operatorname{Log}[b\,x]^{p-1}\,dx$$

- **Program code:**

```
Int[x_^m_.*Sin[a_.*x_^n_.*Log[b_.*x_]^p_.]*Log[b_.*x_]^p_.,x_Symbol] :=
  -Cos[a*x^n*Log[b*x]^p]/(a*n) -
  Dist[p/n,Int[x^m*Sin[a*x^n*Log[b*x]^p]*Log[b*x]^(p-1),x]] /;
FreeQ[{a,b},x] && RationalQ[{m,n,p}] && p>0 && ZeroQ[m-n+1]
```

```
Int[x_^m_.*Cos[a_.*x_^n_.*Log[b_.*x_]^p_.]*Log[b_.*x_]^p_.,x_Symbol] :=
  Sin[a*x^n*Log[b*x]^p]/(a*n) -
  Dist[p/n,Int[x^m*Cos[a*x^n*Log[b*x]^p]*Log[b*x]^(p-1),x]] /;
FreeQ[{a,b},x] && RationalQ[{m,n,p}] && p>0 && ZeroQ[m-n+1]
```

- **Rule: If $p > 0 \land m - n + 1 \neq 0$, then**

$$\begin{aligned}\int x^m \operatorname{Sin}[a\,x^n \operatorname{Log}[b\,x]^p]\,\operatorname{Log}[b\,x]^p\,dx \longrightarrow & -\frac{x^{m-n+1}\operatorname{Cos}[a\,x^n \operatorname{Log}[b\,x]^p]}{a\,n} - \\ & \frac{p}{n}\int x^m \operatorname{Sin}[a\,x^n \operatorname{Log}[b\,x]^p]\,\operatorname{Log}[b\,x]^{p-1}\,dx + \frac{m-n+1}{a\,n}\int x^{m-n}\operatorname{Cos}[a\,x^n \operatorname{Log}[b\,x]^p]\,dx\end{aligned}$$

- **Program code:**

```
Int[x_^m_.*Sin[a_.*x_^n_.*Log[b_.*x_]^p_.]*Log[b_.*x_]^p_.,x_Symbol] :=
  -x^(m-n+1)*Cos[a*x^n*Log[b*x]^p]/(a*n) -
  Dist[p/n,Int[x^m*Sin[a*x^n*Log[b*x]^p]*Log[b*x]^(p-1),x]] +
  Dist[(m-n+1)/(a*n),Int[x^(m-n)*Cos[a*x^n*Log[b*x]^p],x]] /;
FreeQ[{a,b},x] && RationalQ[{m,n,p}] && p>0 && NonzeroQ[m-n+1]
```

```
Int[x_^m_*Cos[a_.*x_^n_.*Log[b_.*x_]^p_.]*Log[b_.*x_]^p_.,x_Symbol] :=
  x^(m-n+1)*Sin[a*x^n*Log[b*x]^p]/(a*n) -
  Dist[p/n,Int[x^m*Cos[a*x^n*Log[b*x]^p]*Log[b*x]^(p-1),x]] -
  Dist[(m-n+1)/(a*n),Int[x^(m-n)*Sin[a*x^n*Log[b*x]^p],x]] /;
FreeQ[{a,b},x] && RationalQ[{m,n,p}] && p>0 && NonzeroQ[m-n+1]
```

$$\int u\,\mathrm{Sin}[a+b\,x]^n\,dx$$

- **Derivation: Algebraic expansion**
- **Basis:** $\mathrm{Sin}[z]^2=\frac{1}{2}-\frac{1}{2}\mathrm{Cos}[2\,z]$
- **Rule: If** $\frac{n-1}{2}\notin\mathbb{Z}$**, then**

$$\int \mathrm{Sin}\left[\frac{a}{2}+\frac{b\,x}{2}\right]^2\mathrm{Sin}[a+b\,x]^n\,dx\;\longrightarrow\;\frac{1}{2}\int \mathrm{Sin}[a+b\,x]^n\,dx-\frac{1}{2}\int \mathrm{Cos}[a+b\,x]\,\mathrm{Sin}[a+b\,x]^n\,dx$$

- **Program code:**

```
Int[Sin[c_.+d_.*x_]^2*Sin[a_.+b_.*x_]^n_.,x_Symbol] :=
  Dist[1/2,Int[Sin[a+b*x]^n,x]] -
  Dist[1/2,Int[Cos[a+b*x]*Sin[a+b*x]^n,x]] /;
FreeQ[{a,b,c,d},x] && ZeroQ[c-a/2] && ZeroQ[d-b/2] && Not[OddQ[n]]
```

- **Derivation: Algebraic expansion**
- **Basis:** $\mathrm{Cos}[z]^2=\frac{1}{2}+\frac{1}{2}\mathrm{Cos}[2\,z]$
- **Rule: If** $\frac{n-1}{2}\notin\mathbb{Z}$**, then**

$$\int \mathrm{Cos}\left[\frac{a}{2}+\frac{b\,x}{2}\right]^2\mathrm{Sin}[a+b\,x]^n\,dx\;\longrightarrow\;\frac{1}{2}\int \mathrm{Sin}[a+b\,x]^n\,dx+\frac{1}{2}\int \mathrm{Cos}[a+b\,x]\,\mathrm{Sin}[a+b\,x]^n\,dx$$

- **Program code:**

```
Int[Cos[c_.+d_.*x_]^2*Sin[a_.+b_.*x_]^n_.,x_Symbol] :=
  Dist[1/2,Int[Sin[a+b*x]^n,x]] +
  Dist[1/2,Int[Cos[a+b*x]*Sin[a+b*x]^n,x]] /;
FreeQ[{a,b,c,d},x] && ZeroQ[c-a/2] && ZeroQ[d-b/2] && Not[OddQ[n]]
```

- **Derivation: Algebraic simplification**
- **Basis:** $\text{Sin}[2z] = 2\,\text{Sin}[z]\,\text{Cos}[z]$
- **Rule: If** $n \in \mathbb{Z}$ **and** u **is a function of trig functions of** $\frac{a}{2}+\frac{b\,x}{2}$**, then**

$$\int u\,\text{Sin}[a+b\,x]^n\,dx \longrightarrow 2^n \int u\,\text{Cos}\left[\frac{a}{2}+\frac{b\,x}{2}\right]^n \text{Sin}\left[\frac{a}{2}+\frac{b\,x}{2}\right]^n dx$$

- **Program code:**

```
Int[u_*Sin[a_.+b_.*x_]^n_.,x_Symbol] :=
  Dist[2^n,Int[u*Cos[a/2+b*x/2]^n*Sin[a/2+b*x/2]^n,x]] /;
FreeQ[{a,b},x] && IntegerQ[n] && ZeroQ[a/2+b*x/2-FunctionOfTrig[u,x]]
```

- **Derivation: Algebraic simplification and piecewise constant extraction**
- **Basis:** $\text{Sin}[2z] = 2\,\text{Sin}[z]\,\text{Cos}[z]$
- **Basis:** $\partial_x \frac{\text{Sin}[a+b\,x]^n}{\text{Sin}\left[\frac{a}{2}+\frac{b\,x}{2}\right]^n \text{Cos}\left[\frac{a}{2}+\frac{b\,x}{2}\right]^n} = 0$
- **Rule: If** $n \in \mathbb{Z}$ **and** u **is a function of trig functions of** $\frac{a}{2}+\frac{b\,x}{2}$**, then**

$$\int u\,\text{Sin}[a+b\,x]^n\,dx \longrightarrow \frac{\text{Sin}[a+b\,x]^n}{\text{Sin}\left[\frac{a}{2}+\frac{b\,x}{2}\right]^n \text{Cos}\left[\frac{a}{2}+\frac{b\,x}{2}\right]^n} \int u\,\text{Cos}\left[\frac{a}{2}+\frac{b\,x}{2}\right]^n \text{Sin}\left[\frac{a}{2}+\frac{b\,x}{2}\right]^n dx$$

- **Program code:**

```
(* Int[u_*Sin[a_.+b_.*x_]^n_,x_Symbol] :=
  Sin[a+b*x]^n/(Sin[a/2+b*x/2]^n*Cos[a/2+b*x/2]^n)*Int[u*Cos[a/2+b*x/2]^n*Sin[a/2+b*x/2]^n,x] /;
FreeQ[{a,b},x] && FractionQ[n] && ZeroQ[a/2+b*x/2-FunctionOfTrig[u,x]] *)
```

$$\int u\,\text{Sin}[v]^2\,dx$$

- **Derivation: Algebraic expansion**

- **Basis:** $\text{Sin}[z]^2 = \frac{1}{2} - \frac{1}{2}\text{Cos}[2\,z]$

- **Rule: If u is a function of trig functions of 2 v, then**

$$\int u\,\text{Sin}[v]^2\,dx \longrightarrow \frac{1}{2}\int u\,dx - \frac{1}{2}\int u\,\text{Cos}[2\,v]\,dx$$

- **Program code:**

```
(* Int[u_*Sin[v_]^2,x_Symbol] :=
  Dist[1/2,Int[u,x]] -
  Dist[1/2,Int[u*Cos[2*v],x]] /;
FunctionOfTrigQ[u,2*v,x] *)
```

```
(* Int[u_*Cos[v_]^2,x_Symbol] :=
  Dist[1/2,Int[u,x]] +
  Dist[1/2,Int[u*Cos[2*v],x]] /;
FunctionOfTrigQ[u,2*v,x] *)
```