$$\int \frac{1}{a + b\,x^n}\,dx$$

- **Reference: G&R 2.124.1a, CRC 60, A&S 3.3.21**
- **Derivation: Primitive rule**
- **Basis:** $\text{ArcTan}'[z] = \frac{1}{1+z^2}$
- **Rule: If** $\frac{a}{b} > 0$**, then**

$$\int \frac{1}{a + b\,x^2}\,dx \longrightarrow \frac{\sqrt{\frac{a}{b}}}{a}\,\text{ArcTan}\left[\frac{x}{\sqrt{\frac{a}{b}}}\right]$$

- **Program code:**

```
Int[1/(a_+b_.*x_^2),x_Symbol] :=
  Rt[a/b,2]/a*ArcTan[x/Rt[a/b,2]] /;
FreeQ[{a,b},x] && PosQ[a/b]
```

- **Reference: G&R 2.124.1b', CRC 61b, A&S 3.3.23**
- **Derivation: Primitive rule**
- **Basis:** $\text{ArcTanh}'[z] = \frac{1}{1-z^2}$
- **Rule: If** $\neg\left(\frac{a}{b} > 0\right)$**, then**

$$\int \frac{1}{a + b\,x^2}\,dx \longrightarrow \frac{\sqrt{-\frac{a}{b}}}{a}\,\text{ArcTanh}\left[\frac{x}{\sqrt{-\frac{a}{b}}}\right]$$

- **Program code:**

```
Int[1/(a_+b_.*x_^2),x_Symbol] :=
  Rt[-a/b,2]/a*ArcTanh[x/Rt[-a/b,2]] /;
FreeQ[{a,b},x] && NegQ[a/b]
```

- **Reference: G&R 2.126.1.2, CRC 74**
- **Derivation: Algebraic expansion**
- **Basis: If** $\frac{r}{s} = \left(\frac{a}{b}\right)^{1/3}$**, then** $\frac{1}{a+b z^3} = \frac{r}{3 a (r+s z)} + \frac{r (2 r-s z)}{3 a \left(r^2-r s z+s^2 z^2\right)}$
- **Rule: If** $\frac{a}{b} > 0$**, let** $\frac{r}{s} = \left(\frac{a}{b}\right)^{1/3}$**, then**

$$\int \frac{1}{a+b x^3} \, dx \longrightarrow \frac{r}{3 a} \int \frac{1}{r+s x} \, dx + \frac{r}{3 a} \int \frac{2 r-s x}{r^2-r s x+s^2 x^2} \, dx$$

- **Program code:**

```
Int[1/(a_+b_.*x_^3),x_Symbol] :=
  Module[{r=Numerator[Rt[a/b,3]], s=Denominator[Rt[a/b,3]]},
  Dist[r/(3*a),Int[1/(r+s*x),x]] +
  Dist[r/(3*a),Int[(2*r-s*x)/(r^2-r*s*x+s^2*x^2),x]]] /;
FreeQ[{a,b},x] && PosQ[a/b]
```

- **Derivation: Algebraic expansion**
- **Basis: If** $\frac{r}{s} = \left(-\frac{a}{b}\right)^{1/3}$**, then** $\frac{1}{a+b z^3} = \frac{r}{3 a (r-s z)} + \frac{r (2 r+s z)}{3 a \left(r^2+r s z+s^2 z^2\right)}$
- **Rule: If** $\neg \left(\frac{a}{b} > 0\right)$**, let** $\frac{r}{s} = \left(-\frac{a}{b}\right)^{1/3}$**, then**

$$\int \frac{1}{a+b x^3} \, dx \longrightarrow \frac{r}{3 a} \int \frac{1}{r-s x} \, dx + \frac{r}{3 a} \int \frac{2 r+s x}{r^2+r s x+s^2 x^2} \, dx$$

- **Program code:**

```
Int[1/(a_+b_.*x_^3),x_Symbol] :=
  Module[{r=Numerator[Rt[-a/b,3]], s=Denominator[Rt[-a/b,3]]},
  Dist[r/(3*a),Int[1/(r-s*x),x]] +
  Dist[r/(3*a),Int[(2*r+s*x)/(r^2+r*s*x+s^2*x^2),x]]] /;
FreeQ[{a,b},x] && NegQ[a/b]
```

- **Reference: G&R 2.132.1.1', CRC 77'**
- **Derivation: Algebraic expansion**
- **Basis: If** $\frac{r}{s} = \left(\frac{a}{b}\right)^{1/4}$**, then** $\frac{1}{a+b z^4} = \frac{r\left(\sqrt{2}\, r-s z\right)}{2\sqrt{2}\, a\left(r^2-\sqrt{2}\, r s z+s^2 z^2\right)} + \frac{r\left(\sqrt{2}\, r+s z\right)}{2\sqrt{2}\, a\left(r^2+\sqrt{2}\, r s z+s^2 z^2\right)}$
- **Rule: If** $\frac{n}{4} \in \mathbb{Z} \bigwedge n > 2 \bigwedge \frac{a}{b} > 0$**, let** $\frac{r}{s} = \left(\frac{a}{b}\right)^{1/4}$**, then**

$$\int \frac{1}{a+b x^n}\, dx \longrightarrow \frac{r}{2\sqrt{2}\, a} \int \frac{\sqrt{2}\, r - s x^{n/4}}{r^2 - \sqrt{2}\, r s x^{n/4} + s^2 x^{n/2}}\, dx + \frac{r}{2\sqrt{2}\, a} \int \frac{\sqrt{2}\, r + s x^{n/4}}{r^2 + \sqrt{2}\, r s x^{n/4} + s^2 x^{n/2}}\, dx$$

- **Program code:**

```
Int[1/(a_+b_.*x_^n_),x_Symbol] :=
  Module[{r=Numerator[Rt[a/b,4]], s=Denominator[Rt[a/b,4]]},
  Dist[r/(2*Sqrt[2]*a),Int[(Sqrt[2]*r-s*x^(n/4))/(r^2-Sqrt[2]*r*s*x^(n/4)+s^2*x^(n/2)),x]] +
  Dist[r/(2*Sqrt[2]*a),Int[(Sqrt[2]*r+s*x^(n/4))/(r^2+Sqrt[2]*r*s*x^(n/4)+s^2*x^(n/2)),x]]] /;
FreeQ[{a,b},x] && IntegerQ[n/4] && n>2 && PositiveQ[a/b]
```

- **Reference: G&R 2.132.1.2', CRC 78'**
- **Derivation: Algebraic expansion**
- **Basis: If** $\frac{r}{s} = \left(-\frac{a}{b}\right)^{1/2}$**, then** $\frac{1}{a+b z^2} = \frac{r}{2a\,(r-s z)} + \frac{r}{2a\,(r+s z)}$
- **Rule: If** $\frac{n}{4} \in \mathbb{Z} \bigwedge n > 2 \bigwedge \neg\left(\frac{a}{b} > 0\right)$**, let** $\frac{r}{s} = \left(-\frac{a}{b}\right)^{1/2}$**, then**

$$\int \frac{1}{a+b x^n}\, dx \longrightarrow \frac{r}{2a} \int \frac{1}{r - s x^{n/2}}\, dx + \frac{r}{2a} \int \frac{1}{r + s x^{n/2}}\, dx$$

- **Program code:**

```
Int[1/(a_+b_.*x_^n_),x_Symbol] :=
  Module[{r=Numerator[Rt[-a/b,2]], s=Denominator[Rt[-a/b,2]]},
  Dist[r/(2*a),Int[1/(r-s*x^(n/2)),x]] +
  Dist[r/(2*a),Int[1/(r+s*x^(n/2)),x]]] /;
FreeQ[{a,b},x] && IntegerQ[n/4] && n>2 && Not[PositiveQ[a/b]]
```

- **Derivation: Algebraic expansion**

- **Basis: If** $\frac{n-2}{4} \in \mathbb{Z}$ **and** $\frac{r}{s} = \left(\frac{a}{b}\right)^{2/n}$**, then** $\frac{1}{a+b z^n} = \frac{2 r}{a n \left(r+s z^2\right)} + \frac{4 r}{a n} \sum_{k=1}^{\frac{n-2}{4}} \frac{r-s \cos\left[\frac{2 (2 k-1) \pi}{n}\right] z^2}{r^2-2 r s \cos\left[\frac{2 (2 k-1) \pi}{n}\right] z^2+s^2 z^4}$

- **Rule: If** $\frac{n-2}{4} \in \mathbb{Z} \bigwedge n > 2 \bigwedge \frac{a}{b} > 0$**, let** $\frac{r}{s} = \left(\frac{a}{b}\right)^{2/n}$**, then**

$$\int \frac{1}{a+b x^n} \, dx \to \frac{2 r}{a n} \int \frac{1}{r+s x^2} \, dx + \frac{4 r}{a n} \int \sum_{k=1}^{\frac{n-2}{4}} \frac{r-s \cos\left[\frac{2 (2 k-1) \pi}{n}\right] x^2}{r^2-2 r s \cos\left[\frac{2 (2 k-1) \pi}{n}\right] x^2+s^2 x^4} \, dx$$

- **Program code:**

```
Int[1/(a_+b_.*x_^n_),x_Symbol] :=
  Module[{r=Numerator[Rt[a/b,n/2]], s=Denominator[Rt[a/b,n/2]]},
  Dist[2*r/(a*n),Int[1/(r+s*x^2),x]] +
  Dist[4*r/(a*n),Int[Sum[(r-s*Cos[2*(2*k-1)*Pi/n]*x^2)/(r^2-2*r*s*Cos[2*(2*k-1)*Pi/n]*x^2+s^2*x^4),
    {k,1,(n-2)/4}],x]]] /;
FreeQ[{a,b},x] && IntegerQ[(n-2)/4] && n>2 && PosQ[a/b]
```

- **Derivation: Algebraic expansion**

- **Basis: If** $\frac{n-2}{4} \in \mathbb{Z}$ **and** $\frac{r}{s} = \left(-\frac{a}{b}\right)^{2/n}$**, then** $\frac{1}{a+b z^n} = \frac{2 r}{a n \left(r-s z^2\right)} + \frac{4 r}{a n} \sum_{k=1}^{\frac{n-2}{4}} \frac{r-s \cos\left[\frac{4 k \pi}{n}\right] z^2}{r^2-2 r s \cos\left[\frac{4 k \pi}{n}\right] z^2+s^2 z^4}$

- **Rule: If** $\frac{n-2}{4} \in \mathbb{Z} \bigwedge n > 2 \bigwedge \neg \left(\frac{a}{b} > 0\right)$**, let** $\frac{r}{s} = \left(-\frac{a}{b}\right)^{2/n}$**, then**

$$\int \frac{1}{a+b x^n} \, dx \to \frac{2 r}{a n} \int \frac{1}{r-s x^2} \, dx + \frac{4 r}{a n} \int \sum_{k=1}^{\frac{n-2}{4}} \frac{r-s \cos\left[\frac{4 k \pi}{n}\right] x^2}{r^2-2 r s \cos\left[\frac{4 k \pi}{n}\right] x^2+s^2 x^4} \, dx$$

- **Program code:**

```
Int[1/(a_+b_.*x_^n_),x_Symbol] :=
  Module[{r=Numerator[Rt[-a/b,n/2]], s=Denominator[Rt[-a/b,n/2]]},
  Dist[2*r/(a*n),Int[1/(r-s*x^2),x]] +
  Dist[4*r/(a*n),Int[Sum[(r-s*Cos[4*k*Pi/n]*x^2)/(r^2-2*r*s*Cos[4*k*Pi/n]*x^2+s^2*x^4),
    {k,1,(n-2)/4}],x]]] /;
FreeQ[{a,b},x] && IntegerQ[(n-2)/4] && n>2 && NegQ[a/b]
```

- **Derivation: Algebraic expansion**

- **Basis: If** $\frac{n-1}{2} \in \mathbb{Z}$ **and** $\frac{r}{s} = \left(\frac{a}{b}\right)^{1/n}$**, then** $\frac{1}{a+b z^n} = \frac{r}{a n (r+s z)} + \frac{2 r}{a n} \sum_{k=1}^{\frac{n-1}{2}} \frac{r-s \operatorname{Cos}\left[\frac{(2 k-1) \pi}{n}\right] z}{r^2-2 r s \operatorname{Cos}\left[\frac{(2 k-1) \pi}{n}\right] z+s^2 z^2}$

- **Rule: If** $\frac{n-1}{2} \in \mathbb{Z} \bigwedge n>1 \bigwedge \frac{a}{b}>0$**, let** $\frac{r}{s} = \left(\frac{a}{b}\right)^{1/n}$**, then**

$$\int \frac{1}{a+b x^n} \, dx \rightarrow \int\left(\frac{r}{a n (r+s x)}+\sum_{k=1}^{\frac{n-1}{2}} \frac{2 r \left(r-s \operatorname{Cos}\left[\frac{(2 k-1) \pi}{n}\right] x\right)}{a n \left(r^2-2 r s \operatorname{Cos}\left[\frac{(2 k-1) \pi}{n}\right] x+s^2 x^2\right)}\right) dx$$

- **Program code:**

```
Int[1/(a_+b_.*x_^n_),x_Symbol] :=
  Module[{r=Numerator[Rt[a/b,n]], s=Denominator[Rt[a/b,n]]},
  Int[r/(a*n*(r+s*x)) +
    Sum[2*r*(r-s*Cos[(2*k-1)*Pi/n]*x)/(a*n*(r^2-2*r*s*Cos[(2*k-1)*Pi/n]*x+s^2*x^2)),
      {k,1,(n-1)/2}],x]] /;
FreeQ[{a,b},x] && OddQ[n] && n>1 && PosQ[a/b]
```

- **Derivation: Algebraic expansion**

- **Basis: If** $\frac{n-1}{2} \in \mathbb{Z}$ **and** $\frac{r}{s} = \left(-\frac{a}{b}\right)^{1/n}$**, then** $\frac{1}{a+b z^n} = \frac{r}{a n (r-s z)} + \frac{2 r}{a n} \sum_{k=1}^{\frac{n-1}{2}} \frac{r+s \operatorname{Cos}\left[\frac{(2 k-1) \pi}{n}\right] z}{r^2+2 r s \operatorname{Cos}\left[\frac{(2 k-1) \pi}{n}\right] z+s^2 z^2}$

- **Rule: If** $\frac{n-1}{2} \in \mathbb{Z} \bigwedge n>1 \bigwedge \neg\left(\frac{a}{b}>0\right)$**, let** $\frac{r}{s} = \left(-\frac{a}{b}\right)^{1/n}$**, then**

$$\int \frac{1}{a+b x^n} \, dx \rightarrow \int\left(\frac{r}{a n (r-s x)}+\sum_{k=1}^{\frac{n-1}{2}} \frac{2 r \left(r+s \operatorname{Cos}\left[\frac{(2 k-1) \pi}{n}\right] x\right)}{a n \left(r^2+2 r s \operatorname{Cos}\left[\frac{(2 k-1) \pi}{n}\right] x+s^2 x^2\right)}\right) dx$$

- **Program code:**

```
Int[1/(a_+b_.*x_^n_),x_Symbol] :=
  Module[{r=Numerator[Rt[-a/b,n]], s=Denominator[Rt[-a/b,n]]},
  Int[r/(a*n*(r-s*x)) +
    Sum[2*r*(r+s*Cos[(2*k-1)*Pi/n]*x)/(a*n*(r^2+2*r*s*Cos[(2*k-1)*Pi/n]*x+s^2*x^2)),
      {k,1,(n-1)/2}],x]] /;
FreeQ[{a,b},x] && OddQ[n] && n>1 && NegQ[a/b]
```

$$\int \frac{x^m}{a + b\,x^n}\,dx$$

- **Reference: G&R 2.126.2, CRC 75**
- **Derivation: Algebraic expansion**
- **Basis: If** $\frac{r}{s} = \left(\frac{a}{b}\right)^{1/3}$**, then** $\frac{z}{a+b\,z^3} = -\frac{r^2}{3\,a\,s\,(r+s\,z)} + \frac{r^2\,(r+s\,z)}{3\,a\,s\,\left(r^2-r\,s\,z+s^2\,z^2\right)}$
- **Rule: If** $\frac{a}{b} > 0$**, let** $\frac{r}{s} = \left(\frac{a}{b}\right)^{1/3}$**, then**

$$\int \frac{x}{a+b\,x^3}\,dx \longrightarrow -\frac{r^2}{3\,a\,s}\int \frac{1}{r+s\,x}\,dx + \frac{r^2}{3\,a\,s}\int \frac{r+s\,x}{r^2-r\,s\,x+s^2\,x^2}\,dx$$

- **Program code:**

```
Int[x_/(a_+b_.*x_^3),x_Symbol] :=
  Module[{r=Numerator[Rt[a/b,3]], s=Denominator[Rt[a/b,3]]},
  Dist[-r^2/(3*a*s),Int[1/(r+s*x),x]] +
  Dist[r^2/(3*a*s),Int[(r+s*x)/(r^2-r*s*x+s^2*x^2),x]]] /;
FreeQ[{a,b},x] && PosQ[a/b]
```

- **Derivation: Algebraic expansion**
- **Basis: If** $\frac{r}{s} = \left(-\frac{a}{b}\right)^{1/3}$**, then** $\frac{1}{a+b\,z^3} = \frac{r^2}{3\,a\,s\,(r-s\,z)} - \frac{r^2\,(r-s\,z)}{3\,a\,s\,\left(r^2+r\,s\,z+s^2\,z^2\right)}$
- **Rule: If** $\neg\left(\frac{a}{b} > 0\right)$**, let** $\frac{r}{s} = \left(-\frac{a}{b}\right)^{1/3}$**, then**

$$\int \frac{x}{a+b\,x^3}\,dx \longrightarrow \frac{r^2}{3\,a\,s}\int \frac{1}{r-s\,x}\,dx - \frac{r^2}{3\,a\,s}\int \frac{r-s\,x}{r^2+r\,s\,x+s^2\,x^2}\,dx$$

- **Program code:**

```
Int[x_/(a_+b_.*x_^3),x_Symbol] :=
  Module[{r=Numerator[Rt[-a/b,3]], s=Denominator[Rt[-a/b,3]]},
  Dist[r^2/(3*a*s),Int[1/(r-s*x),x]] -
  Dist[r^2/(3*a*s),Int[(r-s*x)/(r^2+r*s*x+s^2*x^2),x]]] /;
FreeQ[{a,b},x] && NegQ[a/b]
```

- **Derivation: Integration by substitution**
- **Rule: If** $m, n \in \mathbb{Z} \wedge 0 < m + 1 < n$, **let** $g = \text{GCD}[m + 1, n]$, **if** $g > 1$, **then**

$$\int \frac{x^m}{a + b x^n} \, dx \longrightarrow \frac{1}{g} \, \text{Subst}\left[\int \frac{x^{\frac{m+1}{g}-1}}{a + b x^{n/g}} \, dx, x, x^g\right]$$

- **Program code:**

```
Int[x_^m_./(a_+b_.*x_^n_),x_Symbol] :=
  Module[{g=GCD[m+1,n]},
  Dist[1/g,Subst[Int[x^((m+1)/g-1)/(a+b*x^(n/g)),x],x,x^g]] /;
 g>1] /;
FreeQ[{a,b},x] && IntegersQ[m,n] && 0<m+1<n
```

- **Reference: G&R 2.132.3.1', CRC 81'**
- **Derivation: Algebraic expansion**
- **Basis: If** $\frac{r}{s} = \left(\frac{a}{b}\right)^{1/4}$, **then** $\frac{z^2}{a+b z^4} = \frac{s^3 z}{2\sqrt{2}\, b\, r\left(r^2-\sqrt{2}\, r\, s\, z+s^2 z^2\right)} - \frac{s^3 z}{2\sqrt{2}\, b\, r\left(r^2+\sqrt{2}\, r\, s\, z+s^2 z^2\right)}$
- **Rule: If** $\frac{m}{2} \in \mathbb{Z} \bigwedge m > 0 \bigwedge \frac{a}{b} > 0$, **let** $\frac{r}{s} = \left(\frac{a}{b}\right)^{1/4}$, **then**

$$\int \frac{x^m}{a + b x^{2m}} \, dx \longrightarrow \frac{s^3}{2\sqrt{2}\, b\, r} \int \frac{x^{m/2}}{r^2 - \sqrt{2}\, r\, s\, x^{m/2} + s^2 x^m} \, dx - \frac{s^3}{2\sqrt{2}\, b\, r} \int \frac{x^{m/2}}{r^2 + \sqrt{2}\, r\, s\, x^{m/2} + s^2 x^m} \, dx$$

- **Program code:**

```
Int[x_^m_/(a_+b_.*x_^n_),x_Symbol] :=
  Module[{r=Numerator[Rt[a/b,4]], s=Denominator[Rt[a/b,4]]},
  Dist[s^3/(2*Sqrt[2]*b*r),Int[x^(m/2)/(r^2-Sqrt[2]*r*s*x^(m/2)+s^2*x^m),x]] -
  Dist[s^3/(2*Sqrt[2]*b*r),Int[x^(m/2)/(r^2+Sqrt[2]*r*s*x^(m/2)+s^2*x^m),x]]] /;
FreeQ[{a,b},x] && IntegerQ[m/2] && m>0 && ZeroQ[n-2*m] && PositiveQ[a/b]
```

- **Reference: G&R 2.132.3.2', CRC 82'**
- **Derivation: Algebraic expansion**
- **Basis: If** $\frac{r}{s} = \left(-\frac{a}{b}\right)^{1/2}$**, then** $\frac{z}{a+b z^2} = \frac{s}{2 b (r+s z)} - \frac{s}{2 b (r-s z)}$
- **Rule: If** $\frac{m}{2} \in \mathbb{Z} \bigwedge m > 0 \bigwedge \neg \left(\frac{a}{b} > 0\right)$**, let** $\frac{r}{s} = \left(-\frac{a}{b}\right)^{1/2}$**, then**

$$\int \frac{x^m}{a + b x^{2 m}} \, dx \longrightarrow \frac{s}{2 b} \int \frac{1}{r + s x^m} \, dx - \frac{s}{2 b} \int \frac{1}{r - s x^m} \, dx$$

- **Program code:**

```
Int[x_^m_/(a_+b_.*x_^n_),x_Symbol] :=
  Module[{r=Numerator[Rt[-a/b,2]], s=Denominator[Rt[-a/b,2]]},
  Dist[s/(2*b),Int[1/(r+s*x^m),x]] -
  Dist[s/(2*b),Int[1/(r-s*x^m),x]]] /;
FreeQ[{a,b},x] && EvenQ[m] && m>0 && ZeroQ[n-2*m] && Not[PositiveQ[a/b]]
```

- **Derivation: Algebraic expansion**
- **Basis: If** $\frac{n-2}{4}, \frac{m}{2} \in \mathbb{Z}, 0 \leq m < n$ **and** $\frac{r}{s} = \left(\frac{a}{b}\right)^{2/n}$**, then** $\frac{z^m}{a+b z^n} = -\frac{2 (-r)^{\frac{m}{2}+1}}{a n s^{m/2} (r+s z^2)} + \frac{4 r^{\frac{m}{2}+1}}{a n s^{m/2}} \sum_{k=1}^{\frac{n-2}{4}} \frac{r \operatorname{Cos}\left[\frac{(2 k-1) m \pi}{n}\right] - s \operatorname{Cos}\left[\frac{(2 k-1) (m+2) \pi}{n}\right] z^2}{r^2 - 2 r s \operatorname{Cos}\left[\frac{2 (2 k-1) \pi}{n}\right] z^2 + s^2 z^4}$
- **Rule: If** $\frac{n-2}{4}, \frac{m}{2} \in \mathbb{Z} \bigwedge 0 < m < n \bigwedge \operatorname{CoprimeQ}[m+1, n] \bigwedge \frac{a}{b} > 0$**, let** $\frac{r}{s} = \left(\frac{a}{b}\right)^{2/n}$**, then**

$$\int \frac{x^m}{a + b x^n} \, dx \longrightarrow -\frac{2 (-r)^{\frac{m}{2}+1}}{a n s^{m/2}} \int \frac{1}{r + s x^2} \, dx + \frac{4 r^{\frac{m}{2}+1}}{a n s^{m/2}} \int \sum_{k=1}^{\frac{n-2}{4}} \frac{r \operatorname{Cos}\left[\frac{(2 k-1) m \pi}{n}\right] - s \operatorname{Cos}\left[\frac{(2 k-1) (m+2) \pi}{n}\right] x^2}{r^2 - 2 r s \operatorname{Cos}\left[\frac{2 (2 k-1) \pi}{n}\right] x^2 + s^2 x^4} \, dx$$

- **Program code:**

```
Int[x_^m_./(a_+b_.*x_^n_),x_Symbol] :=
  Module[{r=Numerator[Rt[a/b,n/2]], s=Denominator[Rt[a/b,n/2]]},
  Dist[-2*(-r)^(m/2+1)/(a*n*s^(m/2)),Int[1/(r+s*x^2),x]] +
  Dist[4*r^(m/2+1)/(a*n*s^(m/2)),
    Int[Sum[(r*Cos[(2*k-1)*m*Pi/n]-s*Cos[(2*k-1)*(m+2)*Pi/n]*x^2)/
      (r^2-2*r*s*Cos[2*(2*k-1)*Pi/n]*x^2+s^2*x^4),{k,1,(n-2)/4}],x]]] /;
 FreeQ[{a,b},x] && IntegersQ[(n-2)/4,m/2] && 0<m<n && CoprimeQ[m+1,n] && PosQ[a/b]
```

- **Derivation: Algebraic expansion**

- **Basis:** If $\frac{n-2}{4}, \frac{m}{2} \in \mathbb{Z}, 0 \le m < n$ and $\frac{r}{s} = \left(-\frac{a}{b}\right)^{2/n}$, then $\frac{z^m}{a+b z^n} = \frac{2 r^{\frac{m}{2}+1}}{a n s^{m/2} \left(r-s z^2\right)} + \frac{4 r^{\frac{m}{2}+1}}{a n s^{m/2}} \sum_{k=1}^{\frac{n-2}{4}} \frac{r \operatorname{Cos}\left[\frac{2 k m \pi}{n}\right]-s \operatorname{Cos}\left[\frac{2 k (m+2) \pi}{n}\right] z^2}{r^2-2 r s \operatorname{Cos}\left[\frac{4 k \pi}{n}\right] z^2+s^2 z^4}$

- **Rule:** If $\frac{n-2}{4}, \frac{m}{2} \in \mathbb{Z} \bigwedge 0 < m < n \bigwedge \operatorname{CoprimeQ}[m+1, n] \bigwedge \neg \left(\frac{a}{b} > 0\right)$, let $\frac{r}{s} = \left(-\frac{a}{b}\right)^{2/n}$, then

$$\int \frac{x^m}{a+b x^n} \, dx \longrightarrow \frac{2 r^{\frac{m}{2}+1}}{a n s^{m/2}} \int \frac{1}{r-s x^2} \, dx + \frac{4 r^{\frac{m}{2}+1}}{a n s^{m/2}} \int \sum_{k=1}^{\frac{n-2}{4}} \frac{r \operatorname{Cos}\left[\frac{2 k m \pi}{n}\right]-s \operatorname{Cos}\left[\frac{2 k (m+2) \pi}{n}\right] x^2}{r^2-2 r s \operatorname{Cos}\left[\frac{4 k \pi}{n}\right] x^2+s^2 x^4} \, dx$$

- **Program code:**

```
Int[x_^m_./(a_+b_.*x_^n_),x_Symbol] :=
  Module[{r=Numerator[Rt[-a/b,n/2]], s=Denominator[Rt[-a/b,n/2]]},
  Dist[2*r^(m/2+1)/(a*n*s^(m/2)),Int[1/(r-s*x^2),x]] +
  Dist[4*r^(m/2+1)/(a*n*s^(m/2)),
    Int[Sum[(r*Cos[2*k*m*Pi/n]-s*Cos[2*k*(m+2)*Pi/n]*x^2)/
      (r^2-2*r*s*Cos[4*k*Pi/n]*x^2+s^2*x^4),{k,1,(n-2)/4}],x]]] /;
 FreeQ[{a,b},x] && IntegersQ[(n-2)/4,m/2] && 0<m<n && CoprimeQ[m+1,n] && NegQ[a/b]
```

- **Derivation: Algebraic expansion**

- **Basis:** If $\frac{n-1}{2}, m \in \mathbb{Z}, 0 \le m < n$ and $\frac{r}{s} = \left(\frac{a}{b}\right)^{1/n}$, then $\frac{z^m}{a+b z^n} = -\frac{(-r)^{m+1}}{a n s^m (r+s z)} + \sum_{k=1}^{\frac{n-1}{2}} \frac{2 r^{m+1} \left(r \operatorname{Cos}\left[\frac{(2 k-1) m \pi}{n}\right]-s \operatorname{Cos}\left[\frac{(2 k-1) (m+1) \pi}{n}\right] z\right)}{a n s^m \left(r^2-2 r s \operatorname{Cos}\left[\frac{(2 k-1) \pi}{n}\right] z+s^2 z^2\right)}$

- **Rule:** If $\frac{n-1}{2}, m \in \mathbb{Z} \bigwedge 0 < m < n \bigwedge \operatorname{CoprimeQ}[m+1, n] \bigwedge \frac{a}{b} > 0$, let $\frac{r}{s} = \left(\frac{a}{b}\right)^{1/n}$, then

$$\int \frac{x^m}{a+b x^n} \, dx \longrightarrow \int -\frac{(-r)^{m+1}}{a n s^m (r+s x)} + \sum_{k=1}^{\frac{n-1}{2}} \frac{2 r^{m+1} \left(r \operatorname{Cos}\left[\frac{(2 k-1) m \pi}{n}\right]-s \operatorname{Cos}\left[\frac{(2 k-1) (m+1) \pi}{n}\right] x\right)}{a n s^m \left(r^2-2 r s \operatorname{Cos}\left[\frac{(2 k-1) \pi}{n}\right] x+s^2 x^2\right)} \, dx$$

- **Program code:**

```
Int[x_^m_./(a_+b_.*x_^n_),x_Symbol] :=
  Module[{r=Numerator[Rt[a/b,n]], s=Denominator[Rt[a/b,n]]},
  Int[-(-r)^(m+1)/(a*n*s^m*(r+s*x)) +
    Sum[2*r^(m+1)*(r*Cos[(2*k-1)*m*Pi/n]-s*Cos[(2*k-1)*(m+1)*Pi/n]*x)/
      (a*n*s^m*(r^2-2*r*s*Cos[(2*k-1)*Pi/n]*x+s^2*x^2)),{k,1,(n-1)/2}],x]] /;
FreeQ[{a,b},x] && IntegersQ[(n-1)/2,m] && 0<m<n && CoprimeQ[m+1,n] && PosQ[a/b]
```

- **Derivation: Algebraic expansion**

- **Basis: If** $\frac{n-1}{2}$, $m \in \mathbb{Z}$, $0 \le m < n$ **and** $\frac{r}{s} = \left(-\frac{a}{b}\right)^{1/n}$, **then** $\frac{z^m}{a+b z^n} = \frac{r^{m+1}}{a n s^m (r-s z)} - \sum_{k=1}^{\frac{n-1}{2}} \frac{2 (-r)^{m+1} \left(r \operatorname{Cos}\left[\frac{(2 k-1) m \pi}{n}\right]+s \operatorname{Cos}\left[\frac{(2 k-1) (m+1) \pi}{n}\right] z\right)}{a n s^m \left(r^2+2 r s \operatorname{Cos}\left[\frac{(2 k-1) \pi}{n}\right] z+s^2 z^2\right)}$

- **Rule: If** $\frac{n-1}{2}$, $m \in \mathbb{Z} \bigwedge 0 < m < n \bigwedge \text{CoprimeQ}[m+1, n] \bigwedge \neg \left(\frac{a}{b} > 0\right)$, **let** $\frac{r}{s} = \left(-\frac{a}{b}\right)^{1/n}$, **then**

$$\int \frac{x^m}{a+b x^n} \, dx \longrightarrow \int \frac{r^{m+1}}{a n s^m (r-s x)} - \sum_{k=1}^{\frac{n-1}{2}} \frac{2 (-r)^{m+1} \left(r \operatorname{Cos}\left[\frac{(2 k-1) m \pi}{n}\right] + s \operatorname{Cos}\left[\frac{(2 k-1) (m+1) \pi}{n}\right] x\right)}{a n s^m \left(r^2 + 2 r s \operatorname{Cos}\left[\frac{(2 k-1) \pi}{n}\right] x + s^2 x^2\right)} \, dx$$

- **Program code:**

```
Int[x_^m_./(a_+b_.*x_^n_),x_Symbol] :=
  Module[{r=Numerator[Rt[-a/b,n]], s=Denominator[Rt[-a/b,n]]},
  Int[r^(m+1)/(a*n*s^m*(r-s*x)) -
    Sum[2*(-r)^(m+1)*(r*Cos[(2*k-1)*m*Pi/n]+s*Cos[(2*k-1)*(m+1)*Pi/n]*x)/
      (a*n*s^m*(r^2+2*r*s*Cos[(2*k-1)*Pi/n]*x+s^2*x^2)),{k,1,(n-1)/2}],x]] /;
FreeQ[{a,b},x] && IntegersQ[(n-1)/2,m] && 0<m<n && CoprimeQ[m+1,n] && NegQ[a/b]
```

- **Note: An integration rule for the following algebraic expansion is not needed since** $m+1$ **and** n **are not coprime when m is odd and n even:**

- **Basis: If** $\frac{n}{2}$, $\frac{m+1}{2} \in \mathbb{Z}$, $0 \le m < n$ **and** $\frac{r}{s} = \left(-\frac{a}{b}\right)^{1/n}$, **then** $\frac{z^m}{a+b z^n} = \frac{2 r^{m+1} z}{a n s^{m-1} \left(r^2-s^2 z^2\right)} + \sum_{k=1}^{\frac{n}{2}-1} \frac{2 r^{m+1} \left(r \operatorname{Cos}\left[\frac{2 k m \pi}{n}\right]-s \operatorname{Cos}\left[\frac{2 k (m+1) \pi}{n}\right] z\right)}{a n s^m \left(r^2-2 r s \operatorname{Cos}\left[\frac{2 k \pi}{n}\right] z+s^2 z^2\right)}$

$$\int \frac{c + d\,x^m}{a + b\,x^{2\,m}}\,dx$$

- **Derivation: Algebraic expansion**
- **Basis: If** $\frac{r}{s} = \left(\frac{a}{b}\right)^{1/4}$**, then** $\frac{c+d\,z^m}{a+b\,z^{2\,m}} = \frac{r\left(\sqrt{2}\,c\,r\,s+\left(c\,s^2-d\,r^2\right)z^{m/2}\right)}{2\sqrt{2}\,a\,s\left(r^2+\sqrt{2}\,r\,s\,z^{m/2}+s^2\,z^m\right)} + \frac{r\left(\sqrt{2}\,c\,r\,s-\left(c\,s^2-d\,r^2\right)z^{m/2}\right)}{2\sqrt{2}\,a\,s\left(r^2-\sqrt{2}\,r\,s\,z^{m/2}+s^2\,z^m\right)}$
- **Rule: If** $\frac{m}{2} \in \mathbb{Z} \bigwedge m > 0 \bigwedge \frac{a}{b} > 0$**, let** $\frac{r}{s} = \left(\frac{a}{b}\right)^{1/4}$**, then**

$$\int \frac{c+d\,x^m}{a+b\,x^{2\,m}}\,dx \to \frac{r}{2\sqrt{2}\,a\,s}\int \frac{\sqrt{2}\,c\,r\,s+\left(c\,s^2-d\,r^2\right)x^{m/2}}{r^2+\sqrt{2}\,r\,s\,x^{m/2}+s^2\,x^m}\,dx + \frac{r}{2\sqrt{2}\,a\,s}\int \frac{\sqrt{2}\,c\,r\,s-\left(c\,s^2-d\,r^2\right)x^{m/2}}{r^2-\sqrt{2}\,r\,s\,x^{m/2}+s^2\,x^m}\,dx$$

- **Program code:**

```
Int[(c_.+d_.*x_^m_)/(a_+b_.*x_^n_),x_Symbol] :=
  Module[{r=Numerator[Rt[a/b,4]], s=Denominator[Rt[a/b,4]]},
  Dist[r/(2*Sqrt[2]*a*s),
    Int[(Sqrt[2]*c*r*s+(c*s^2-d*r^2)*x^(m/2))/(r^2+Sqrt[2]*r*s*x^(m/2)+s^2*x^m),x]] +
  Dist[r/(2*Sqrt[2]*a*s),
    Int[(Sqrt[2]*c*r*s-(c*s^2-d*r^2)*x^(m/2))/(r^2-Sqrt[2]*r*s*x^(m/2)+s^2*x^m),x]]] /;
FreeQ[{a,b,c,d},x] && IntegerQ[m/2] && m>0 && ZeroQ[n-2*m] && PosQ[a/b]
```

- **Derivation: Algebraic expansion**
- **Basis: If** $\frac{r}{s} = \left(-\frac{a}{b}\right)^{1/2}$**, then** $\frac{c+d\,z^m}{a+b\,z^{2\,m}} = \frac{c\,s+d\,r}{2\,(a\,s+b\,r\,z^m)} + \frac{c\,s-d\,r}{2\,(a\,s-b\,r\,z^m)}$
- **Rule: If** $\frac{m}{2} \in \mathbb{Z} \bigwedge m > 0 \bigwedge \neg\left(\frac{a}{b} > 0\right) \bigwedge b\,c^2 + a\,d^2 \neq 0$**, let** $\frac{r}{s} = \left(-\frac{a}{b}\right)^{1/2}$**, then**

$$\int \frac{c+d\,x^m}{a+b\,x^{2\,m}}\,dx \to \frac{c\,s+d\,r}{2}\int \frac{1}{a\,s+b\,r\,x^m}\,dx + \frac{c\,s-d\,r}{2}\int \frac{1}{a\,s-b\,r\,x^m}\,dx$$

- **Program code:**

```
Int[(c_.+d_.*x_^m_)/(a_+b_.*x_^n_),x_Symbol] :=
  Module[{r=Numerator[Rt[-a/b,2]], s=Denominator[Rt[-a/b,2]]},
  Dist[(c*s+d*r)/2, Int[1/(a*s+b*r*x^m),x]] +
  Dist[(c*s-d*r)/2, Int[1/(a*s-b*r*x^m),x]]] /;
FreeQ[{a,b,c,d},x] && IntegerQ[m/2] && m>0 && ZeroQ[n-2*m] && NegQ[a/b] && NonzeroQ[b*c^2+a*d^2]
```

$$\int (a + b\,x^n)^p \, dx$$

- **Reference: G&R 2.110.2, CRC 88d special case when $n\,(p+1)+1=0$**
- **Rule: If $n\,(p+1)+1=0$, then**

$$\int (a + b\,x^n)^p \, dx \longrightarrow \frac{x\,(a+b\,x^n)^{p+1}}{a}$$

- **Program code:**

```
Int[(a_+b_.*x_^n_)^p_,x_Symbol] :=
  x*(a+b*x^n)^(p+1)/a /;
FreeQ[{a,b,n,p},x] && ZeroQ[n*(p+1)+1]
```

- **Reference: G&R 2.110.2, CRC 88d**
- **Derivation: Integration by parts**
- **Basis:** $(a+b\,x^n)^p = x^{n\,(p+1)+1}\,\frac{(a+b\,x^n)^p}{x^{n\,(p+1)+1}}$
- **Basis:** $\int \frac{(a+b\,x^n)^p}{x^{n\,(p+1)+1}}\,dx = -\frac{(a+b\,x^n)^{p+1}}{x^{n\,(p+1)}\,a\,n\,(p+1)}$
- **Note: Requirement that $n>1$ ensures new term is a proper fraction.**
- **Rule: If $n,\ p \in \mathbb{Z} \wedge n>1 \wedge p<-1$, then**

$$\int (a + b\,x^n)^p \, dx \longrightarrow -\frac{x\,(a+b\,x^n)^{p+1}}{a\,n\,(p+1)} + \frac{n\,(p+1)+1}{a\,n\,(p+1)} \int (a+b\,x^n)^{p+1}\,dx$$

- **Program code:**

```
Int[(a_+b_.*x_^n_)^p_,x_Symbol] :=
  -x*(a+b*x^n)^(p+1)/(a*n*(p+1)) +
  Dist[(n*(p+1)+1)/(a*n*(p+1)),Int[(a+b*x^n)^(p+1),x]] /;
FreeQ[{a,b},x] && IntegersQ[n,p] && n>1 && p<-1
```

$$\int x^m (a + b x^n)^p \, dx$$

- **Reference: G&R 2.110.6, CRC 88c special case when $m + n (p + 1) + 1 = 0$**

- **Rule: If $m + n (p + 1) + 1 = 0 \land m + 1 \neq 0 \land p \neq -2$, then**

$$\int x^m (a + b x^n)^p \, dx \longrightarrow \frac{x^{m+1} (a + b x^n)^{p+1}}{a (m + 1)}$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  x^(m+1)*(a+b*x^n)^(p+1)/(a*(m+1)) /;
FreeQ[{a,b,m,n,p},x] && ZeroQ[m+n*(p+1)+1] && NonzeroQ[m+1] && NonzeroQ[p+2]
```

- **Reference: G&R 2.110.4**

- **Derivation: Integration by parts**

- **Basis: $x^m (a + b x^n)^p = x^{m-n+1} (a + b x^n)^p x^{n-1}$**

- **Basis: $\int (a + b x^n)^p x^{n-1} \, dx = \frac{(a+b x^n)^{p+1}}{b n (p+1)}$**

- **Note: Requirement that $m < 2 n - 1$ ensures new term is a proper fraction.**

- **Rule: If $m, n, p \in \mathbb{Z} \land 1 < n \leq m < 2 n - 1 \land p < -1$, then**

$$\int x^m (a + b x^n)^p \, dx \longrightarrow \frac{x^{m-n+1} (a + b x^n)^{p+1}}{b n (p + 1)} - \frac{m - n + 1}{b n (p + 1)} \int x^{m-n} (a + b x^n)^{p+1} \, dx$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  x^(m-n+1)*(a+b*x^n)^(p+1)/(b*n*(p+1)) -
  Dist[(m-n+1)/(b*n*(p+1)),Int[x^(m-n)*(a+b*x^n)^(p+1),x]] /;
FreeQ[{a,b},x] && IntegersQ[m,n,p] && 1<n<=m<2*n-1 && p<-1
```

- **Reference: G&R 2.110.2, CRC 88d**
- **Derivation: Integration by parts**
- **Basis:** $x^m (a+b x^n)^p = x^{m+n (p+1)+1} \frac{(a+b x^n)^p}{x^{n (p+1)+1}}$
- **Basis:** $\int \frac{(a+b x^n)^p}{x^{n (p+1)+1}} dx = -\frac{(a+b x^n)^{p+1}}{x^{n (p+1)} a n (p+1)}$
- **Note: Requirement that $m+1<n$ ensures new term is a proper fraction.**
- **Rule: If $m, n, p \in \mathbb{Z} \wedge n>1 \wedge 0<m+1<n \wedge p<-1 \wedge m+n (p+1)+1 \neq 0$, then**

$$\int x^m (a+b x^n)^p dx \longrightarrow -\frac{x^{m+1} (a+b x^n)^{p+1}}{a n (p+1)}+\frac{m+n (p+1)+1}{a n (p+1)} \int x^m (a+b x^n)^{p+1} dx$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  -x^(m+1)*(a+b*x^n)^(p+1)/(a*n*(p+1)) +
  Dist[(m+n*(p+1)+1)/(a*n*(p+1)),Int[x^m*(a+b*x^n)^(p+1),x]] /;
FreeQ[{a,b},x] && IntegersQ[m,n,p] && n>1 && 0<m+1<n && p<-1 && NonzeroQ[m+n*(p+1)+1]
```

- **Reference: G&R 2.110.6, CRC 88c**
- **Derivation: Integration by parts**
- **Basis:** $x^m (a+b x^n)^p = \frac{x^m}{(a+b x^n)^{\frac{m+n+1}{n}}} (a+b x^n)^{\frac{m+n (p+1)+1}{n}}$
- **Basis:** $\int \frac{x^m}{(a+b x^n)^{\frac{m+n+1}{n}}} dx = \frac{x^{m+1}}{(a+b x^n)^{\frac{m+1}{n}} (a (m+1))}$
- **Note: Requirement that $m+1<n$ ensures new term is a proper fraction.**
- **Rule: If $m, n, p, \frac{m+n (p+1)+1}{n} \in \mathbb{Z} \bigwedge m<-1 \bigwedge n>0 \bigwedge 0<n-2 (m+n (p+1)+1)<n p$, then**

$$\int x^m (a+b x^n)^p dx \longrightarrow \frac{x^{m+1} (a+b x^n)^{p+1}}{a (m+1)}-\frac{b (m+n (p+1)+1)}{a (m+1)} \int x^{m+n} (a+b x^n)^p dx$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  x^(m+1)*(a+b*x^n)^(p+1)/(a*(m+1)) -
  Dist[b*(m+n*(p+1)+1)/(a*(m+1)),Int[x^(m+n)*(a+b*x^n)^p,x]] /;
FreeQ[{a,b},x] && IntegersQ[m,n,p,(m+n*(p+1)+1)/n] && m<-1 && n>0 &&
0<n-2(m+n*(p+1)+1)<n*p
```

- **Reference: G&R 2.110.5, CRC 88a**
- **Derivation: Inverted integration by parts**
- **Rule: If** $m, n, p, \frac{m+1}{n} \in \mathbb{Z} \bigwedge m+n\,p+1 \neq 0 \bigwedge \frac{m+1}{n} > 0 \bigwedge \frac{2\,m}{n} < p+1 \bigwedge 0 < n \leq m$**, then**

$$\int x^m\,(a+b\,x^n)^p\,dx \longrightarrow \frac{x^{m-n+1}\,(a+b\,x^n)^{p+1}}{b\,(m+n\,p+1)} - \frac{a\,(m-n+1)}{b\,(m+n\,p+1)}\int x^{m-n}\,(a+b\,x^n)^p\,dx$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  x^(m-n+1)*(a+b*x^n)^(p+1)/(b*(m+n*p+1)) -
  Dist[a*(m-n+1)/(b*(m+n*p+1)),Int[x^(m-n)*(a+b*x^n)^p,x]] /;
FreeQ[{a,b},x] && IntegersQ[m,n,p,(m+1)/n] && NonzeroQ[m+n*p+1] &&
(m+1)/n>0 && 2*m/n<p+1 && 0<n<=m
```

- **Derivation: Algebraic expansion**
- **Program code:**

```
Int[x_^m_.*(a_+b_.*x_^n_.)^p_.,x_Symbol] :=
  Int[Expand[x^m*(a+b*x^n)^p],x] /;
FreeQ[{a,b,m,n},x] && IntegerQ[p] && p>0 && ExpandIntegrandQ[m,n,p]
```

- **Reference: G&R 2.110.4**
- **Derivation: Integration by parts**
- **Basis:** $x^m\,(a+b\,x^n)^p = x^{m-n+1}\,(a+b\,x^n)^p\,x^{n-1}$
- **Note: Requirement that** $m < 2\,n-1$ **ensures new term is a proper fraction.**
- **Note: Unfortunately this rule is necessary to prevent the Ostrogradskiy-Hermite method from being applied instead of substituting for** $c+d\,x$**.**
- **Rule: If** $m, n, p \in \mathbb{Z} \wedge n > 1 \wedge p < -1 \wedge n \leq m < 2\,n-1$**, then**

$$\int (c+d\,x)^m\,(a+b\,(c+d\,x)^n)^p\,dx \longrightarrow$$
$$\frac{(c+d\,x)^{m-n+1}\,(a+b\,(c+d\,x)^n)^{p+1}}{b\,d\,n\,(p+1)} - \frac{m-n+1}{b\,n\,(p+1)}\int (c+d\,x)^{m-n}\,(a+b\,(c+d\,x)^n)^{p+1}\,dx$$

- **Program code:**

```
Int[(c_+d_.*x_)^m_.*(a_+b_.*(c_+d_.*x_)^n_)^p_,x_Symbol] :=
  (c+d*x)^(m-n+1)*(a+b*(c+d*x)^n)^(p+1)/(b*d*n*(p+1)) -
  Dist[(m-n+1)/(b*n*(p+1)),Int[(c+d*x)^(m-n)*(a+b*(c+d*x)^n)^(p+1),x]] /;
FreeQ[{a,b,c,d},x] && IntegersQ[m,n,p] && n>1 && p<-1 && n<=m<2*n-1
```

$$\int \frac{(a + b\, x^n)^m}{b + \frac{a}{x^n}}\, dx$$

- **Derivation: Algebraic simplification**
- **Basis:** $\frac{a+b\, x^n}{b+\frac{a}{x^n}} = x^n$
- **Rule:**

$$\int \frac{(a + b\, x^n)^m}{b + \frac{a}{x^n}}\, dx \longrightarrow \int x^n\, (a + b\, x^n)^{m-1}\, dx$$

- **Program code:**

```
Int[(a_+b_.*x_^n_.)^m_/(b_+a_.*x_^p_.),x_Symbol] :=
  Int[x^n*(a+b*x^n)^(m-1), x] /;
FreeQ[{a,b,m,n,p},x] && ZeroQ[n+p]
```

$$\int (a\,x^p + b\,x^q)^n\,dx$$

- **Derivation: Algebraic simplification**
- **Basis:** $a\,z^p + b\,z^q = z^p\,(a + b\,z^{q-p})$
- **Rule: If** $n \in \mathbb{Z}$, **then**

$$\int (a\,x^p + b\,x^q)^n\,dx \longrightarrow \int x^{n\,p}\,(a + b\,x^{q-p})^n\,dx$$

- **Program code:**

```
Int[(a_.*x_^p_.+b_.*x_^q_.)^n_,x_Symbol] :=
  Int[x^(n*p)*(a+b*x^(q-p))^n,x] /;
FreeQ[{a,b,p,q},x] && IntegerQ[n] && Not[FractionQ[p]] && Not[FractionQ[q]] && Not[NegativeQ[q-p]]
```

- **Derivation: Algebraic simplification**
- **Basis:** $a\,z^p + b\,z^q = z^p\,(a + b\,z^{q-p})$
- **Rule: If** $n \in \mathbb{Z}$, **then**

$$\int x^m\,(a\,x^p + b\,x^q)^n\,dx \longrightarrow \int x^{m+n\,p}\,(a + b\,x^{q-p})^n\,dx$$

- **Program code:**

```
Int[x_^m_.*(a_.*x_^p_.+b_.*x_^q_.)^n_,x_Symbol] :=
  Int[x^(m+n*p)*(a+b*x^(q-p))^n,x] /;
FreeQ[{a,b,m,p,q},x] && IntegerQ[n] &&
Not[FractionQ[p]] && Not[FractionQ[q]] && Not[FractionQ[m]] && Not[NegativeQ[q-p]]
```