$$\int \mathrm{ArcSec}[a+b\,x]^n\,dx$$

- **Reference: G&R 2.821.2, CRC 445', A&S 4.4.62'**
- **Derivation: Integration by parts**
- **Rule:**

$$\int \mathrm{ArcSec}[a+b\,x]\,dx \longrightarrow \frac{(a+b\,x)\,\mathrm{ArcSec}[a+b\,x]}{b}-\int \frac{1}{(a+b\,x)\sqrt{1-\frac{1}{(a+b\,x)^2}}}\,dx$$

- **Program code:**

```
Int[ArcSec[a_.+b_.*x_],x_Symbol] :=
  (a+b*x)*ArcSec[a+b*x]/b -
  Int[1/((a+b*x)*Sqrt[1-1/(a+b*x)^2]),x] /;
FreeQ[{a,b},x]
```

$$\int x^m \operatorname{ArcSec}[a + b x] \, dx$$

- **Derivation: Integration by substitution**
- **Rule: If** $m \in \mathbb{Z} \land m > 0$, **then**

$$\int x^m \operatorname{ArcSec}[a + b x] \, dx \longrightarrow \frac{1}{b} \operatorname{Subst}\left[\int \left(-\frac{a}{b} + \frac{x}{b}\right)^m \operatorname{ArcSec}[x] \, dx,\ x,\ a + b x\right]$$

- **Program code:**

```
Int[x_^m_.*ArcSec[a_+b_.*x_],x_Symbol] :=
  Dist[1/b,Subst[Int[(-a/b+x/b)^m*ArcSec[x],x],x,a+b*x]] /;
FreeQ[{a,b},x] && IntegerQ[m] && m>0
```

- **Reference: CRC 474**
- **Derivation: Integration by parts**
- **Rule: If** $m + 1 \neq 0$, **then**

$$\int x^m \operatorname{ArcSec}[a x] \, dx \longrightarrow \frac{x^{m+1} \operatorname{ArcSec}[a x]}{m + 1} - \frac{1}{a (m + 1)} \int \frac{x^{m-1}}{\sqrt{1 - \frac{1}{a^2 x^2}}} \, dx$$

- **Program code:**

```
Int[x_^m_.*ArcSec[a_.*x_],x_Symbol] :=
  x^(m+1)*ArcSec[a*x]/(m+1) -
  Dist[1/(a*(m+1)),Int[x^(m-1)/Sqrt[1-1/(a*x)^2],x]] /;
FreeQ[{a,m},x] && NonzeroQ[m+1]
```

- **Reference: CRC 474**
- **Derivation: Integration by parts**
- **Rule: If $m+1 \neq 0$, then**

$$\int x^m \operatorname{ArcSec}[a+b\,x]\,dx \longrightarrow \frac{x^{m+1}\operatorname{ArcSec}[a+b\,x]}{m+1}-\frac{b}{m+1}\int\frac{x^{m+1}}{(a+b\,x)^2\sqrt{1-\frac{1}{(a+b\,x)^2}}}\,dx$$

- **Program code:**

```
Int[x_^m_.*ArcSec[a_.+b_.*x_],x_Symbol] :=
  x^(m+1)*ArcSec[a+b*x]/(m+1) -
  Dist[b/(m+1),Int[x^(m+1)/((a+b*x)^2*Sqrt[1-1/(a+b*x)^2]),x]] /;
FreeQ[{a,b,m},x] && NonzeroQ[m+1]
```

```
(* Int[ArcSec[a_.*x_^n_.]/x_,x_Symbol] :=
  I*ArcSec[a*x^n]^2/(2*n) -
  ArcSec[a*x^n]*Log[1-1/(I/(x^n*a)+Sqrt[1-1/(x^(2*n)*a^2)])^2]/n +
  I*PolyLog[2,1/(I/(x^n*a)+Sqrt[1-1/(x^(2*n)*a^2)])^2]/(2*n) /;
(* Sqrt[-1/a^2]*a*ArcCsc[a*x^n]^2/(2*n) +
  Pi*Log[x]/2 -
  Sqrt[-1/a^2]*a*ArcSinh[Sqrt[-1/a^2]/x^n]*Log[1-1/(Sqrt[-(1/a^2)]/x^n+Sqrt[1-1/(x^(2*n)*a^2)])^2]/n +
  Sqrt[-1/a^2]*a*PolyLog[2, 1/(Sqrt[-1/a^2]/x^n+Sqrt[1-1/(x^(2*n)*a^2)])^2]/(2*n) *)
FreeQ[{a,n},x] *)
```

$$\int u\,\text{ArcSec}\left[\frac{c}{a+b\,x^n}\right]^m dx$$

- **Derivation: Algebraic simplification**
- **Basis:** $\text{ArcSec}[z] = \text{ArcCos}\left[\frac{1}{z}\right]$
- **Rule:**

$$\int u\,\text{ArcSec}\left[\frac{c}{a+b\,x^n}\right]^m dx \longrightarrow \int u\,\text{ArcCos}\left[\frac{a}{c}+\frac{b\,x^n}{c}\right]^m dx$$

- **Program code:**

```
Int[u_.*ArcSec[c_./(a_.+b_.*x_^n_.)]^m_.,x_Symbol] :=
  Int[u*ArcCos[a/c+b*x^n/c]^m,x] /;
FreeQ[{a,b,c,n,m},x]
```

$$\int \text{ArcSec}[u]\, dx$$

- **Derivation: Integration by parts**
- **Rule: If u is free of inverse functions, then**

$$\int \text{ArcSec}[u]\, dx \longrightarrow x\, \text{ArcSec}[u] - \int \frac{x\, \partial_x u}{u^2 \sqrt{1-\frac{1}{u^2}}}\, dx$$

- **Program code:**

```
Int[ArcSec[u_],x_Symbol] :=
  x*ArcSec[u] -
  Int[Regularize[x*D[u,x]/(u^2*Sqrt[1-1/u^2]),x],x] /;
InverseFunctionFreeQ[u,x] && Not[FunctionOfExponentialOfLinear[u,x]]
```