$$\int \text{SinhIntegral}[a+b\,x]^n\,dx$$

- **Derivation: Integration by parts**
- **Rule:**

$$\int \text{SinhIntegral}[a+b\,x]\,dx \longrightarrow \frac{(a+b\,x)\,\text{SinhIntegral}[a+b\,x]}{b}-\frac{\text{Cosh}[a+b\,x]}{b}$$

- **Program code:**

```
Int[SinhIntegral[a_.+b_.*x_],x_Symbol] :=
  (a+b*x)*SinhIntegral[a+b*x]/b - Cosh[a+b*x]/b/;
FreeQ[{a,b},x]
```

```
Int[CoshIntegral[a_.+b_.*x_],x_Symbol] :=
  (a+b*x)*CoshIntegral[a+b*x]/b - Sinh[a+b*x]/b /;
FreeQ[{a,b},x]
```

- **Derivation: Integration by parts**
- **Rule:**

$$\int \text{SinhIntegral}[a+b\,x]^2\,dx \longrightarrow$$
$$\frac{(a+b\,x)\,\text{SinhIntegral}[a+b\,x]^2}{b}-2\int \text{Sinh}[a+b\,x]\,\text{SinhIntegral}[a+b\,x]\,dx$$

- **Program code:**

```
Int[SinhIntegral[a_.+b_.*x_]^2,x_Symbol] :=
  (a+b*x)*SinhIntegral[a+b*x]^2/b -
  Dist[2,Int[Sinh[a+b*x]*SinhIntegral[a+b*x],x]] /;
FreeQ[{a,b},x]
```

```
Int[CoshIntegral[a_.+b_.*x_]^2,x_Symbol] :=
  (a+b*x)*CoshIntegral[a+b*x]^2/b -
  Dist[2,Int[Cosh[a+b*x]*CoshIntegral[a+b*x],x]] /;
FreeQ[{a,b},x]
```

$$\int x^m \,\text{SinhIntegral}[a+b\,x]^n \,dx$$

- **Derivation: Integration by parts**
- **Rule: If $m+1 \neq 0$, then**

$$\int x^m \,\text{SinhIntegral}[a+b\,x]\,dx \longrightarrow \frac{x^{m+1}\,\text{SinhIntegral}[a+b\,x]}{m+1} - \frac{b}{m+1}\int \frac{x^{m+1}\,\text{Sinh}[a+b\,x]}{a+b\,x}\,dx$$

- **Program code:**

```
Int[x_^m_.*SinhIntegral[a_.+b_.*x_],x_Symbol] :=
  x^(m+1)*SinhIntegral[a+b*x]/(m+1) -
  Dist[b/(m+1),Int[x^(m+1)*Sinh[a+b*x]/(a+b*x),x]] /;
FreeQ[{a,b,m},x] && NonzeroQ[m+1]
```

```
Int[x_^m_.*CoshIntegral[a_.+b_.*x_],x_Symbol] :=
  x^(m+1)*CoshIntegral[a+b*x]/(m+1) -
  Dist[b/(m+1),Int[x^(m+1)*Cosh[a+b*x]/(a+b*x),x]] /;
FreeQ[{a,b,m},x] && NonzeroQ[m+1]
```

- **Derivation: Integration by parts**
- **Rule: If $m \in \mathbb{Z} \land m > 0$, then**

$$\int x^m \,\text{SinhIntegral}[b\,x]^2\,dx \longrightarrow \frac{x^{m+1}\,\text{SinhIntegral}[b\,x]^2}{m+1} - \frac{2}{m+1}\int x^m\,\text{Sinh}[b\,x]\,\text{SinhIntegral}[b\,x]\,dx$$

- **Program code:**

```
Int[x_^m_.*SinhIntegral[b_.*x_]^2,x_Symbol] :=
  x^(m+1)*SinhIntegral[b*x]^2/(m+1) -
  Dist[2/(m+1),Int[x^m*Sinh[b*x]*SinhIntegral[b*x],x]] /;
FreeQ[b,x] && IntegerQ[m] && m>0
```

```
Int[x_^m_.*CoshIntegral[b_.*x_]^2,x_Symbol] :=
  x^(m+1)*CoshIntegral[b*x]^2/(m+1) -
  Dist[2/(m+1),Int[x^m*Cosh[b*x]*CoshIntegral[b*x],x]] /;
FreeQ[b,x] && IntegerQ[m] && m>0
```

- **Derivation: Iterated integration by parts**
- **Rule: If $m \in \mathbb{Z} \land m > 0$, then**

$$\int x^m \text{SinhIntegral}[a+b\,x]^2\,dx \longrightarrow \frac{x^{m+1}\,\text{SinhIntegral}[a+b\,x]^2}{m+1}+\frac{a\,x^m\,\text{SinhIntegral}[a+b\,x]^2}{b\,(m+1)}-\frac{2}{m+1}\int x^m \,\text{Sinh}[a+b\,x]\,\text{SinhIntegral}[a+b\,x]\,dx-\frac{a\,m}{b\,(m+1)}\int x^{m-1}\,\text{SinhIntegral}[a+b\,x]^2\,dx$$

- **Program code:**

```
Int[x_^m_.*SinhIntegral[a_+b_.*x_]^2,x_Symbol] :=
  x^(m+1)*SinhIntegral[a+b*x]^2/(m+1) +
  a*x^m*SinhIntegral[a+b*x]^2/(b*(m+1)) -
  Dist[2/(m+1),Int[x^m*Sinh[a+b*x]*SinhIntegral[a+b*x],x]] -
  Dist[a*m/(b*(m+1)),Int[x^(m-1)*SinhIntegral[a+b*x]^2,x]] /;
FreeQ[{a,b},x] && IntegerQ[m] && m>0
```

```
Int[x_^m_.*CoshIntegral[a_+b_.*x_]^2,x_Symbol] :=
  x^(m+1)*CoshIntegral[a+b*x]^2/(m+1) +
  a*x^m*CoshIntegral[a+b*x]^2/(b*(m+1)) -
  Dist[2/(m+1),Int[x^m*Cosh[a+b*x]*CoshIntegral[a+b*x],x]] -
  Dist[a*m/(b*(m+1)),Int[x^(m-1)*CoshIntegral[a+b*x]^2,x]] /;
FreeQ[{a,b},x] && IntegerQ[m] && m>0
```

- **Derivation: Inverted integration by parts**
- **Rule: If $m \in \mathbb{Z} \land m < -2$, then**

$$\int x^m \text{SinhIntegral}[a+b\,x]^2\,dx \longrightarrow \frac{b\,x^{m+2}\,\text{SinhIntegral}[a+b\,x]^2}{a\,(m+1)}+\frac{x^{m+1}\,\text{SinhIntegral}[a+b\,x]^2}{m+1}-\frac{2\,b}{a\,(m+1)}\int x^{m+1}\,\text{Sinh}[a+b\,x]\,\text{SinhIntegral}[a+b\,x]\,dx-\frac{b\,(m+2)}{a\,(m+1)}\int x^{m+1}\,\text{SinhIntegral}[a+b\,x]^2\,dx$$

- **Program code:**

```
(* Int[x_^m_.*SinhIntegral[a_+b_.*x_]^2,x_Symbol] :=
  b*x^(m+2)*SinhIntegral[a+b*x]^2/(a*(m+1)) +
  x^(m+1)*SinhIntegral[a+b*x]^2/(m+1) -
  Dist[2*b/(a*(m+1)),Int[x^(m+1)*Sinh[a+b*x]*SinhIntegral[a+b*x],x]] -
  Dist[b*(m+2)/(a*(m+1)),Int[x^(m+1)*SinhIntegral[a+b*x]^2,x]] /;
FreeQ[{a,b},x] && IntegerQ[m] && m<-2 *)
```

```
(* Int[x_^m_.*CoshIntegral[a_+b_.*x_]^2,x_Symbol] :=
  b*x^(m+2)*CoshIntegral[a+b*x]^2/(a*(m+1)) +
  x^(m+1)*CoshIntegral[a+b*x]^2/(m+1) -
  Dist[2*b/(a*(m+1)),Int[x^(m+1)*Cosh[a+b*x]*CoshIntegral[a+b*x],x]] -
  Dist[b*(m+2)/(a*(m+1)),Int[x^(m+1)*CoshIntegral[a+b*x]^2,x]] /;
FreeQ[{a,b},x] && IntegerQ[m] && m<-2 *)
```

$$\int \mathrm{Sinh}[a+b\,x]\ \mathrm{SinhIntegral}[c+d\,x]\,dx$$

- **Derivation: Integration by parts**

- **Rule:**

$$\int \mathrm{Sinh}[a+b\,x]\ \mathrm{SinhIntegral}[c+d\,x]\,dx \longrightarrow \frac{\mathrm{Cosh}[a+b\,x]\ \mathrm{SinhIntegral}[c+d\,x]}{b}-\frac{d}{b}\int\frac{\mathrm{Cosh}[a+b\,x]\ \mathrm{Sinh}[c+d\,x]}{c+d\,x}\,dx$$

- **Program code:**

```
Int[Sinh[a_.+b_.*x_]*SinhIntegral[c_.+d_.*x_],x_Symbol] :=
  Cosh[a+b*x]*SinhIntegral[c+d*x]/b -
  Dist[d/b,Int[Cosh[a+b*x]*Sinh[c+d*x]/(c+d*x),x]] /;
FreeQ[{a,b,c,d},x]
```

```
Int[Cosh[a_.+b_.*x_]*CoshIntegral[c_.+d_.*x_],x_Symbol] :=
  Sinh[a+b*x]*CoshIntegral[c+d*x]/b -
  Dist[d/b,Int[Sinh[a+b*x]*Cosh[c+d*x]/(c+d*x),x]] /;
FreeQ[{a,b,c,d},x]
```

$\int x^m \operatorname{Sinh}[a+b\,x]\operatorname{SinhIntegral}[c+d\,x]\,dx$

- **Derivation: Integration by parts**
- **Rule: If $m\in\mathbb{Z}\land m>0$, then**

$$\int x^m \operatorname{Sinh}[a+b\,x]\operatorname{SinhIntegral}[c+d\,x]\,dx \longrightarrow \frac{x^m \operatorname{Cosh}[a+b\,x]\operatorname{SinhIntegral}[c+d\,x]}{b}-\frac{d}{b}\int\frac{x^m \operatorname{Cosh}[a+b\,x]\operatorname{Sinh}[c+d\,x]}{c+d\,x}\,dx-\frac{m}{b}\int x^{m-1}\operatorname{Cosh}[a+b\,x]\operatorname{SinhIntegral}[c+d\,x]\,dx$$

- **Program code:**

```
Int[x_^m_.*Sinh[a_.+b_.*x_]*SinhIntegral[c_.+d_.*x_],x_Symbol] :=
  x^m*Cosh[a+b*x]*SinhIntegral[c+d*x]/b -
  Dist[d/b,Int[x^m*Cosh[a+b*x]*Sinh[c+d*x]/(c+d*x),x]] -
  Dist[m/b,Int[x^(m-1)*Cosh[a+b*x]*SinhIntegral[c+d*x],x]] /;
FreeQ[{a,b,c,d},x] && IntegerQ[m] && m>0
```

```
Int[x_^m_.*Cosh[a_.+b_.*x_]*CoshIntegral[c_.+d_.*x_],x_Symbol] :=
  x^m*Sinh[a+b*x]*CoshIntegral[c+d*x]/b -
  Dist[d/b,Int[x^m*Sinh[a+b*x]*Cosh[c+d*x]/(c+d*x),x]] -
  Dist[m/b,Int[x^(m-1)*Sinh[a+b*x]*CoshIntegral[c+d*x],x]] /;
FreeQ[{a,b,c,d},x] && IntegerQ[m] && m>0
```

- **Derivation: Inverted integration by parts**
- **Rule: If $m\in\mathbb{Z}\land m<-1$, then**

$$\int x^m \operatorname{Sinh}[a+b\,x]\operatorname{SinhIntegral}[c+d\,x]\,dx \longrightarrow \frac{x^{m+1} \operatorname{Sinh}[a+b\,x]\operatorname{SinhIntegral}[c+d\,x]}{m+1}-\frac{d}{m+1}\int\frac{x^{m+1} \operatorname{Sinh}[a+b\,x]\operatorname{Sinh}[c+d\,x]}{c+d\,x}\,dx-\frac{b}{m+1}\int x^{m+1}\operatorname{Cosh}[a+b\,x]\operatorname{SinhIntegral}[c+d\,x]\,dx$$

- **Program code:**

```
Int[x_^m_*Sinh[a_.+b_.*x_]*SinhIntegral[c_.+d_.*x_],x_Symbol] :=
  x^(m+1)*Sinh[a+b*x]*SinhIntegral[c+d*x]/(m+1) -
  Dist[d/(m+1),Int[x^(m+1)*Sinh[a+b*x]*Sinh[c+d*x]/(c+d*x),x]] -
  Dist[b/(m+1),Int[x^(m+1)*Cosh[a+b*x]*SinhIntegral[c+d*x],x]] /;
FreeQ[{a,b,c,d},x] && IntegerQ[m] && m<-1
```

```
Int[x_^m_.*Cosh[a_.+b_.*x_]*CoshIntegral[c_.+d_.*x_],x_Symbol] :=
  x^(m+1)*Cosh[a+b*x]*CoshIntegral[c+d*x]/(m+1) -
  Dist[d/(m+1),Int[x^(m+1)*Cosh[a+b*x]*Cosh[c+d*x]/(c+d*x),x]] -
  Dist[b/(m+1),Int[x^(m+1)*Sinh[a+b*x]*CoshIntegral[c+d*x],x]] /;
FreeQ[{a,b,c,d},x] && IntegerQ[m] && m<-1
```

$\int \text{Cosh}[a+b\,x]\ \text{SinhIntegral}[c+d\,x]\,dx$

- **Derivation: Integration by parts**
- **Rule:**

$$\int \text{Cosh}[a+b\,x]\ \text{SinhIntegral}[c+d\,x]\,dx \longrightarrow \frac{\text{Sinh}[a+b\,x]\ \text{SinhIntegral}[c+d\,x]}{b}-\frac{d}{b}\int\frac{\text{Sinh}[a+b\,x]\ \text{Sinh}[c+d\,x]}{c+d\,x}\,dx$$

- **Program code:**

```
Int[Cosh[a_.+b_.*x_]*SinhIntegral[c_.+d_.*x_],x_Symbol] :=
  Sinh[a+b*x]*SinhIntegral[c+d*x]/b -
  Dist[d/b,Int[Sinh[a+b*x]*Sinh[c+d*x]/(c+d*x),x]] /;
FreeQ[{a,b,c,d},x]
```

```
Int[Sinh[a_.+b_.*x_]*CoshIntegral[c_.+d_.*x_],x_Symbol] :=
  Cosh[a+b*x]*CoshIntegral[c+d*x]/b -
  Dist[d/b,Int[Cosh[a+b*x]*Cosh[c+d*x]/(c+d*x),x]] /;
FreeQ[{a,b,c,d},x]
```

$\int x^m \operatorname{Cosh}[a+b\,x]\operatorname{SinhIntegral}[c+d\,x]\,dx$

- **Derivation: Integration by parts**
- **Rule: If $m\in\mathbb{Z}\land m>0$, then**

$$\int x^m \operatorname{Cosh}[a+b\,x]\operatorname{SinhIntegral}[c+d\,x]\,dx \longrightarrow \frac{x^m \operatorname{Sinh}[a+b\,x]\operatorname{SinhIntegral}[c+d\,x]}{b}-\frac{d}{b}\int\frac{x^m \operatorname{Sinh}[a+b\,x]\operatorname{Sinh}[c+d\,x]}{c+d\,x}\,dx-\frac{m}{b}\int x^{m-1}\operatorname{Sinh}[a+b\,x]\operatorname{SinhIntegral}[c+d\,x]\,dx$$

- **Program code:**

```
Int[x_^m_.*Cosh[a_.+b_.*x_]*SinhIntegral[c_.+d_.*x_],x_Symbol] :=
  x^m*Sinh[a+b*x]*SinhIntegral[c+d*x]/b -
  Dist[d/b,Int[x^m*Sinh[a+b*x]*Sinh[c+d*x]/(c+d*x),x]] -
  Dist[m/b,Int[x^(m-1)*Sinh[a+b*x]*SinhIntegral[c+d*x],x]] /;
FreeQ[{a,b,c,d},x] && IntegerQ[m] && m>0
```

```
Int[x_^m_.*Sinh[a_.+b_.*x_]*CoshIntegral[c_.+d_.*x_],x_Symbol] :=
  x^m*Cosh[a+b*x]*CoshIntegral[c+d*x]/b -
  Dist[d/b,Int[x^m*Cosh[a+b*x]*Cosh[c+d*x]/(c+d*x),x]] -
  Dist[m/b,Int[x^(m-1)*Cosh[a+b*x]*CoshIntegral[c+d*x],x]] /;
FreeQ[{a,b,c,d},x] && IntegerQ[m] && m>0
```

- **Derivation: Inverted integration by parts**
- **Rule: If $m\in\mathbb{Z}\land m<-1$, then**

$$\int x^m \operatorname{Cosh}[a+b\,x]\operatorname{SinhIntegral}[c+d\,x]\,dx \longrightarrow \frac{x^{m+1} \operatorname{Cosh}[a+b\,x]\operatorname{SinhIntegral}[c+d\,x]}{m+1}-\frac{d}{m+1}\int\frac{x^{m+1} \operatorname{Cosh}[a+b\,x]\operatorname{Sinh}[c+d\,x]}{c+d\,x}\,dx-\frac{b}{m+1}\int x^{m+1}\operatorname{Sinh}[a+b\,x]\operatorname{SinhIntegral}[c+d\,x]\,dx$$

- **Program code:**

```
Int[x_^m_.*Cosh[a_.+b_.*x_]*SinhIntegral[c_.+d_.*x_],x_Symbol] :=
  x^(m+1)*Cosh[a+b*x]*SinhIntegral[c+d*x]/(m+1) -
  Dist[d/(m+1),Int[x^(m+1)*Cosh[a+b*x]*Sinh[c+d*x]/(c+d*x),x]] -
  Dist[b/(m+1),Int[x^(m+1)*Sinh[a+b*x]*SinhIntegral[c+d*x],x]] /;
FreeQ[{a,b,c,d},x] && IntegerQ[m] && m<-1
```

```
Int[x_^m_*Sinh[a_.+b_.*x_]*CoshIntegral[c_.+d_.*x_],x_Symbol] :=
  x^(m+1)*Sinh[a+b*x]*CoshIntegral[c+d*x]/(m+1) -
  Dist[d/(m+1),Int[x^(m+1)*Sinh[a+b*x]*Cosh[c+d*x]/(c+d*x),x]] -
  Dist[b/(m+1),Int[x^(m+1)*Cosh[a+b*x]*CoshIntegral[c+d*x],x]] /;
FreeQ[{a,b,c,d},x] && IntegerQ[m] && m<-1
```