$\int x^m \text{Sec}[a+b x]^n \, dx$

- **Derivation: Integration by parts**
- **Rule: If $m \in \mathbb{Z} \land m > 0$, then**

$$\int x^m \text{Sec}[a+b x] \, dx \longrightarrow -\frac{2 i x^m \text{ArcTan}\left[e^{i a+i b x}\right]}{b} + \frac{2 i m}{b} \int x^{m-1} \text{ArcTan}\left[e^{i a+i b x}\right] dx$$

- **Program code:**

```
Int[x_^m_.*Sec[a_.+b_.*x_],x_Symbol] :=
  -2*I*x^m*ArcTan[E^(I*a+I*b*x)]/b +
  Dist[2*I*m/b,Int[x^(m-1)*ArcTan[E^(I*a+I*b*x)],x]] /;
FreeQ[{a,b},x] && IntegerQ[m] && m>0
```

```
Int[x_^m_.*Csc[a_.+b_.*x_],x_Symbol] :=
  -2*x^m*ArcTanh[E^(I*a+I*b*x)]/b +
  Dist[2*m/b,Int[x^(m-1)*ArcTanh[E^(I*a+I*b*x)],x]] /;
FreeQ[{a,b},x] && IntegerQ[m] && m>0
```

- **Reference: CRC 430, A&S 4.3.125**
- **Rule: If $m > 0$, then**

$$\int x^m \text{Sec}[a+b x]^2 \, dx \longrightarrow \frac{x^m \text{Tan}[a+b x]}{b} - \frac{m}{b} \int x^{m-1} \text{Tan}[a+b x] \, dx$$

- **Program code:**

```
Int[x_^m_.*Sec[a_.+b_.*x_]^2,x_Symbol] :=
  x^m*Tan[a+b*x]/b -
  Dist[m/b,Int[x^(m-1)*Tan[a+b*x],x]] /;
FreeQ[{a,b},x] && RationalQ[m] && m>0
```

- **Reference: CRC 428, A&S 4.3.121**

```
Int[x_^m_.*Csc[a_.+b_.*x_]^2,x_Symbol] :=
  -x^m*Cot[a+b*x]/b +
  Dist[m/b,Int[x^(m-1)*Cot[a+b*x],x]] /;
FreeQ[{a,b},x] && RationalQ[m] && m>0
```

- **Reference: G&R 2.643.2 special case when $m = 1$, CRC 431, A&S 4.3.126**
- **Rule: If $n > 1 \wedge n \neq 2$, then**

$$\int x \operatorname{Sec}[a+b x]^n \, dx \longrightarrow \frac{x \operatorname{Tan}[a+b x] \operatorname{Sec}[a+b x]^{n-2}}{b (n-1)} - \frac{\operatorname{Sec}[a+b x]^{n-2}}{b^2 (n-1) (n-2)} + \frac{n-2}{n-1} \int x \operatorname{Sec}[a+b x]^{n-2} \, dx$$

- **Program code:**

```
Int[x_*Sec[a_.+b_.*x_]^n_,x_Symbol] :=
  x*Tan[a+b*x]*Sec[a+b*x]^(n-2)/(b*(n-1)) -
  Sec[a+b*x]^(n-2)/(b^2*(n-1)*(n-2)) +
  Dist[(n-2)/(n-1),Int[x*Sec[a+b*x]^(n-2),x]] /;
FreeQ[{a,b},x] && RationalQ[n] && n>1 && n≠2
```

- **Reference: G&R 2.643.1 special case when $m = 1$, CRC 429', A&S 4.3.122**

```
Int[x_*Csc[a_.+b_.*x_]^n_,x_Symbol] :=
  -x*Cot[a+b*x]*Csc[a+b*x]^(n-2)/(b*(n-1)) -
  Csc[a+b*x]^(n-2)/(b^2*(n-1)*(n-2)) +
  Dist[(n-2)/(n-1),Int[x*Csc[a+b*x]^(n-2),x]] /;
FreeQ[{a,b},x] && RationalQ[n] && n>1 && n≠2
```

- **Reference: G&R 2.643.2**
- **Rule: If $m > 1 \wedge n > 1 \wedge n \neq 2$, then**

$$\int x^m \operatorname{Sec}[a+b x]^n \, dx \longrightarrow \frac{x^m \operatorname{Tan}[a+b x] \operatorname{Sec}[a+b x]^{n-2}}{b (n-1)} - \frac{m x^{m-1} \operatorname{Sec}[a+b x]^{n-2}}{b^2 (n-1) (n-2)} + \frac{n-2}{n-1} \int x^m \operatorname{Sec}[a+b x]^{n-2} \, dx + \frac{m (m-1)}{b^2 (n-1) (n-2)} \int x^{m-2} \operatorname{Sec}[a+b x]^{n-2} \, dx$$

- **Program code:**

```
Int[x_^m_*Sec[a_.+b_.*x_]^n_,x_Symbol] :=
  x^m*Tan[a+b*x]*Sec[a+b*x]^(n-2)/(b*(n-1)) -
  m*x^(m-1)*Sec[a+b*x]^(n-2)/(b^2*(n-1)*(n-2)) +
  Dist[(n-2)/(n-1),Int[x^m*Sec[a+b*x]^(n-2),x]] +
  Dist[m*(m-1)/(b^2*(n-1)*(n-2)),Int[x^(m-2)*Sec[a+b*x]^(n-2),x]] /;
FreeQ[{a,b},x] && RationalQ[{m,n}] && m>1 && n>1 && n≠2
```

- **Reference: G&R 2.643.1**

```
Int[x_^m_*Csc[a_.+b_.*x_]^n_,x_Symbol] :=
  -x^m*Cot[a+b*x]*Csc[a+b*x]^(n-2)/(b*(n-1)) -
  m*x^(m-1)*Csc[a+b*x]^(n-2)/(b^2*(n-1)*(n-2)) +
  Dist[(n-2)/(n-1),Int[x^m*Csc[a+b*x]^(n-2),x]] +
  Dist[m*(m-1)/(b^2*(n-1)*(n-2)),Int[x^(m-2)*Csc[a+b*x]^(n-2),x]] /;
FreeQ[{a,b},x] && RationalQ[{m,n}] && m>1 && n>1 && n≠2
```

- **Reference: G&R 2.631.3 special case when $m = 1$**

- **Rule: If $n < -1$, then**

$$\int x \operatorname{Sec}[a+b x]^n \, dx \longrightarrow \frac{\operatorname{Sec}[a+b x]^n}{b^2 n^2} - \frac{x \operatorname{Sin}[a+b x] \operatorname{Sec}[a+b x]^{n+1}}{b n} + \frac{n+1}{n} \int x \operatorname{Sec}[a+b x]^{n+2} \, dx$$

- **Program code:**

```
Int[x_*Sec[a_.+b_.*x_]^n_,x_Symbol] :=
  Sec[a+b*x]^n/(b^2*n^2) -
  x*Sin[a+b*x]*Sec[a+b*x]^(n+1)/(b*n) +
  Dist[(n+1)/n,Int[x*Sec[a+b*x]^(n+2),x]] /;
FreeQ[{a,b},x] && RationalQ[n] && n<-1
```

- **Reference: G&R 2.631.2 special case when $m = 1$**

```
Int[x_*Csc[a_.+b_.*x_]^n_,x_Symbol] :=
  Csc[a+b*x]^n/(b^2*n^2) +
  x*Cos[a+b*x]*Csc[a+b*x]^(n+1)/(b*n) +
  Dist[(n+1)/n,Int[x*Csc[a+b*x]^(n+2),x]] /;
FreeQ[{a,b},x] && RationalQ[n] && n<-1
```

- **Reference: G&R 2.631.3**

- **Rule: If $m > 1 \land n < -1$, then**

$$\int x^m \operatorname{Sec}[a+b x]^n \, dx \longrightarrow \frac{m x^{m-1} \operatorname{Sec}[a+b x]^n}{b^2 n^2} - \frac{x^m \operatorname{Sin}[a+b x] \operatorname{Sec}[a+b x]^{n+1}}{b n} + \frac{n+1}{n} \int x^m \operatorname{Sec}[a+b x]^{n+2} \, dx - \frac{m (m-1)}{b^2 n^2} \int x^{m-2} \operatorname{Sec}[a+b x]^n \, dx$$

- **Program code:**

```
Int[x_^m_*Sec[a_.+b_.*x_]^n_,x_Symbol] :=
  m*x^(m-1)*Sec[a+b*x]^n/(b^2*n^2) -
  x^m*Sin[a+b*x]*Sec[a+b*x]^(n+1)/(b*n) +
  Dist[(n+1)/n,Int[x^m*Sec[a+b*x]^(n+2),x]] -
  Dist[m*(m-1)/(b^2*n^2),Int[x^(m-2)*Sec[a+b*x]^n,x]] /;
FreeQ[{a,b},x] && RationalQ[{m,n}] && m>1 && n<-1
```

- **Reference: G&R 2.631.2**

```
Int[x_^m_*Csc[a_.+b_.*x_]^n_,x_Symbol] :=
  m*x^(m-1)*Csc[a+b*x]^n/(b^2*n^2) +
  x^m*Cos[a+b*x]*Csc[a+b*x]^(n+1)/(b*n) +
  Dist[(n+1)/n,Int[x^m*Csc[a+b*x]^(n+2),x]] -
  Dist[m*(m-1)/(b^2*n^2),Int[x^(m-2)*Csc[a+b*x]^n,x]] /;
FreeQ[{a,b},x] && RationalQ[{m,n}] && m>1 && n<-1
```

$$\int (a + b \operatorname{Sec}[c + d x]^n)^m \, dx$$

- **Derivation: Algebraic simplification**
- **Basis: If** $a + b = 0$, **then** $a + b \operatorname{Sec}[z]^2 = b \operatorname{Tan}[z]^2$
- **Rule: If** $a + b = 0 \land m \in \mathbb{Z}$, **then**

$$\int u \left(a + b \operatorname{Sec}[v]^2\right)^m dx \longrightarrow b^m \int u \operatorname{Tan}[v]^{2m} dx$$

- **Program code:**

```
Int[u_.*(a_+b_.*Sec[v_]^2)^m_,x_Symbol] :=
  Dist[b^m,Int[u*Tan[v]^(2*m),x]] /;
FreeQ[{a,b,m},x] && ZeroQ[a+b] && IntegerQ[m]
```

```
Int[u_.*(a_+b_.*Csc[v_]^2)^m_,x_Symbol] :=
  Dist[b^m,Int[u*Cot[v]^(2*m),x]] /;
FreeQ[{a,b,m},x] && ZeroQ[a+b] && IntegerQ[m]
```

- **Derivation: Algebraic simplification**
- **Basis: If** $a + b = 0$, **then** $a + b \operatorname{Sec}[z]^2 = b \operatorname{Tan}[z]^2$
- **Rule: If** $a + b = 0 \land m \notin \mathbb{Z}$, **then**

$$\int u \left(a + b \operatorname{Sec}[v]^2\right)^m dx \longrightarrow \int u \left(b \operatorname{Tan}[v]^2\right)^m dx$$

- **Program code:**

```
Int[u_.*(a_+b_.*Sec[v_]^2)^m_,x_Symbol] :=
  Int[u*(b*Tan[v]^2)^m,x] /;
FreeQ[{a,b,m},x] && ZeroQ[a+b] && Not[IntegerQ[m]]
```

```
Int[u_.*(a_+b_.*Csc[v_]^2)^m_,x_Symbol] :=
  Int[u*(b*Cot[v]^2)^m,x] /;
FreeQ[{a,b,m},x] && ZeroQ[a+b] && Not[IntegerQ[m]]
```

- **Derivation: Algebraic simplification**
- **Basis: If** $n \in \mathbb{Z}$, **then** $a + b\,\text{Sec}[z]^n = \frac{b+a\,\text{Cos}[z]^n}{\text{Cos}[z]^n}$
- **Rule: If** $m,\ n \in \mathbb{Z} \bigwedge m < 0 \bigwedge n > 1$, **then**

$$\int (a + b\,\text{Sec}[v]^n)^m\,dx \longrightarrow \int \frac{(b + a\,\text{Cos}[v]^n)^m}{\text{Cos}[v]^{m\,n}}\,dx$$

- **Program code:**

```
Int[(a_+b_.*Sec[v_]^n_)^m_,x_Symbol] :=
  Int[(b+a*Cos[v]^n)^m/Cos[v]^(m*n),x] /;
FreeQ[{a,b},x] && IntegersQ[m,n] && m<0 && n>1
```

```
Int[(a_+b_.*Csc[v_]^n_)^m_,x_Symbol] :=
  Int[(b+a*Sin[v]^n)^m/Sin[v]^(m*n),x] /;
FreeQ[{a,b},x] && IntegersQ[m,n] && m<0 && n>1
```

- **Derivation: Algebraic simplification**
- **Basis: If** $n \in \mathbb{Z}$, **then** $a + b\,\text{Sec}[z]^n = \frac{b+a\,\text{Cos}[z]^n}{\text{Cos}[z]^n}$
- **Rule: If** $m,\ n,\ p \in \mathbb{Z} \bigwedge m < 0 \bigwedge n > 0$, **then**

$$\int \text{Cos}[v]^p\,(a + b\,\text{Sec}[v]^n)^m\,dx \longrightarrow \int \text{Cos}[v]^{p-m\,n}\,(b + a\,\text{Cos}[v]^n)^m\,dx$$

- **Program code:**

```
(* Int[Cos[v_]^p_.*(a_+b_.*Sec[v_]^n_.)^m_,x_Symbol] :=
  Int[Cos[v]^(p-m*n)*(b+a*Cos[v]^n)^m,x] /;
FreeQ[{a,b},x] && IntegersQ[m,n,p] && m<0 && n>0 *)
```

```
(* Int[Sin[v_]^p_.*(a_+b_.*Csc[v_]^n_.)^m_,x_Symbol] :=
  Int[Sin[v]^(p-m*n)*(b+a*Sin[v]^n)^m,x] /;
FreeQ[{a,b},x] && IntegersQ[m,n,p] && m<0 && n>0 *)
```

$$\int \mathrm{Csc}[a+b\,x]^m\,\mathrm{Sec}[a+b\,x]^n\,dx$$

- **Reference: G&R 2.526.49, CRC 329**
- **Rule: If** $b>0$**, then**

$$\int \mathrm{Csc}[a+b\,x]\,\mathrm{Sec}[a+b\,x]\,dx \longrightarrow \frac{\mathrm{Log}[\mathrm{Tan}[a+b\,x]]}{b}$$

- **Program code:**

```
Int[Csc[a_.+b_.*x_]*Sec[a_.+b_.*x_],x_Symbol] :=
  Log[Tan[a+b*x]]/b /;
FreeQ[{a,b},x] && PosQ[b]
```

- **Rule: If** $m+n-2=0\wedge n-1\neq 0\wedge n>0$**, then**

$$\int \mathrm{Csc}[a+b\,x]^m\,\mathrm{Sec}[a+b\,x]^n\,dx \longrightarrow \frac{\mathrm{Csc}[a+b\,x]^{m-1}\,\mathrm{Sec}[a+b\,x]^{n-1}}{b\,(n-1)}$$

- **Program code:**

```
Int[Csc[a_.+b_.*x_]^m_*Sec[a_.+b_.*x_]^n_,x_Symbol] :=
  Csc[a+b*x]^(m-1)*Sec[a+b*x]^(n-1)/(b*(n-1)) /;
FreeQ[{a,b,m,n},x] && ZeroQ[m+n-2] && NonzeroQ[n-1] && PosQ[n]
```

- **Derivation: Integration by substitution**
- **Basis: If** $m,\ n,\ \frac{m+n}{2}\in\mathbb{Z}$**, then** $\mathrm{Csc}[z]^m\,\mathrm{Sec}[z]^n = \frac{\left(1+\mathrm{Tan}[z]^2\right)^{\frac{m+n}{2}-1}}{\mathrm{Tan}[z]^m}\,\mathrm{Tan}'[z]$
- **Rule: If** $m,\ n,\ \frac{m+n}{2}\in\mathbb{Z}\wedge 0<m\leq n$**, then**

$$\int \mathrm{Csc}[a+b\,x]^m\,\mathrm{Sec}[a+b\,x]^n\,dx \longrightarrow \frac{1}{b}\,\mathrm{Subst}\left[\int \frac{\left(1+x^2\right)^{\frac{m+n}{2}-1}}{x^m}\,dx,\ x,\ \mathrm{Tan}[a+b\,x]\right]$$

- **Program code:**

```
Int[Csc[a_.+b_.*x_]^m_.*Sec[a_.+b_.*x_]^n_,x_Symbol] :=
  Dist[1/b,Subst[Int[Regularize[(1+x^2)^((m+n)/2-1)/x^m,x],x],x,Tan[a+b*x]]] /;
FreeQ[{a,b},x] && IntegersQ[m,n] && EvenQ[m+n] && 0<m<=n
```

- **Reference: G&R 2.510.4**
- **Rule: If $m < -1 \land n > 1$, then**

$$\int Csc[a+b\,x]^m\,Sec[a+b\,x]^n\,dx \longrightarrow \frac{Csc[a+b\,x]^{m+1}\,Sec[a+b\,x]^{n-1}}{b\,(n-1)}+\frac{m+1}{n-1}\int Csc[a+b\,x]^{m+2}\,Sec[a+b\,x]^{n-2}\,dx$$

- **Program code:**

```
Int[Csc[a_.+b_.*x_]^m_*Sec[a_.+b_.*x_]^n_,x_Symbol] :=
  Csc[a+b*x]^(m+1)*Sec[a+b*x]^(n-1)/(b*(n-1)) +
  Dist[(m+1)/(n-1),Int[Csc[a+b*x]^(m+2)*Sec[a+b*x]^(n-2),x]] /;
FreeQ[{a,b},x] && RationalQ[{m,n}] && m<-1 && n>1
```

- **Reference: G&R 2.510.6, CRC 334b, A&S 4.3.128a**
- **Rule: If $n > 1 \bigwedge \frac{m+n}{2} \notin \mathbb{Z} \bigwedge \neg\left(\frac{n}{2},\ \frac{m-1}{2} \in \mathbb{Z} \bigwedge m > 1\right)$, then**

$$\int Csc[a+b\,x]^m\,Sec[a+b\,x]^n\,dx \longrightarrow \frac{Csc[a+b\,x]^{m-1}\,Sec[a+b\,x]^{n-1}}{b\,(n-1)}+\frac{m+n-2}{n-1}\int Csc[a+b\,x]^{m}\,Sec[a+b\,x]^{n-2}\,dx$$

- **Program code:**

```
Int[Csc[a_.+b_.*x_]^m_.*Sec[a_.+b_.*x_]^n_,x_Symbol] :=
  Csc[a+b*x]^(m-1)*Sec[a+b*x]^(n-1)/(b*(n-1)) +
  Dist[(m+n-2)/(n-1),Int[Csc[a+b*x]^m*Sec[a+b*x]^(n-2),x]] /;
FreeQ[{a,b,m},x] && RationalQ[n] && n>1 && Not[EvenQ[m+n]] && Not[EvenQ[n] && OddQ[m] && m>1]
```

- **Reference: G&R 2.510.5, CRC 323a, A&S 4.3.127a**
- **Rule: If $n < -1 \land m + n \neq 0$, then**

$$\int Csc[a+b\,x]^m\,Sec[a+b\,x]^n\,dx \longrightarrow -\frac{Csc[a+b\,x]^{m-1}\,Sec[a+b\,x]^{n+1}}{b\,(m+n)}+\frac{n+1}{m+n}\int Csc[a+b\,x]^{m}\,Sec[a+b\,x]^{n+2}\,dx$$

- **Program code:**

```
Int[Csc[a_.+b_.*x_]^m_.*Sec[a_.+b_.*x_]^n_,x_Symbol] :=
  -Csc[a+b*x]^(m-1)*Sec[a+b*x]^(n+1)/(b*(m+n)) +
  Dist[(n+1)/(m+n),Int[Csc[a+b*x]^m*Sec[a+b*x]^(n+2),x]] /;
FreeQ[{a,b,m},x] && RationalQ[n] && n<-1 && NonzeroQ[m+n]
```

$$\int \mathrm{Csc}[a+b\,x]^m\,\mathrm{Sec}[a+b\,x]^n\,dx$$

- **Reference: G&R 2.526.49', CRC 329'**
- **Rule: If** $\neg\,(b>0)$**, then**

$$\int \mathrm{Csc}[a+b\,x]\,\mathrm{Sec}[a+b\,x]\,dx \longrightarrow -\frac{\mathrm{Log}[\mathrm{Cot}[a+b\,x]]}{b}$$

- **Program code:**

```
Int[Csc[a_.+b_.*x_]*Sec[a_.+b_.*x_],x_Symbol] :=
  -Log[Cot[a+b*x]]/b /;
FreeQ[{a,b},x] && NegQ[b]
```

- **Rule: If** $m+n-2=0\,\wedge\,m-1\neq 0\,\wedge\,m>0$**, then**

$$\int \mathrm{Csc}[a+b\,x]^m\,\mathrm{Sec}[a+b\,x]^n\,dx \longrightarrow -\frac{\mathrm{Csc}[a+b\,x]^{m-1}\,\mathrm{Sec}[a+b\,x]^{n-1}}{b\,(m-1)}$$

- **Program code:**

```
Int[Csc[a_.+b_.*x_]^m_*Sec[a_.+b_.*x_]^n_,x_Symbol] :=
  -Csc[a+b*x]^(m-1)*Sec[a+b*x]^(n-1)/(b*(m-1)) /;
FreeQ[{a,b,m,n},x] && ZeroQ[m+n-2] && NonzeroQ[m-1] && PosQ[m]
```

- **Derivation: Integration by substitution**
- **Basis: If** $m,\ n,\ \frac{m+n}{2}\in\mathbb{Z}$**, then** $\mathrm{Csc}[z]^m\,\mathrm{Sec}[z]^n = -\frac{\left(1+\mathrm{Cot}[z]^2\right)^{\frac{m+n}{2}-1}}{\mathrm{Cot}[z]^n}\,\mathrm{Cot}'[z]$
- **Rule: If** $m,\ n,\ \frac{m+n}{2}\in\mathbb{Z}\,\bigwedge\,0<n<m$**, then**

$$\int \mathrm{Csc}[a+b\,x]^m\,\mathrm{Sec}[a+b\,x]^n\,dx \longrightarrow -\frac{1}{b}\,\mathrm{Subst}\Big[\mathrm{Int}\Big[\frac{\left(1+x^2\right)^{\frac{m+n}{2}-1}}{x^n},\,x\Big],\,x,\,\mathrm{Cot}[a+b\,x]\Big]$$

- **Program code:**

```
Int[Csc[a_.+b_.*x_]^m_*Sec[a_.+b_.*x_]^n_.,x_Symbol] :=
  Dist[-1/b,Subst[Int[Regularize[(1+x^2)^((m+n)/2-1)/x^n,x],x],x,Cot[a+b*x]]] /;
FreeQ[{a,b},x] && IntegersQ[m,n] && EvenQ[m+n] && 0<n<m
```

- **Reference: G&R 2.510.1**

- **Rule: If $m > 1 \land n < -1$, then**

$$\int \mathrm{Csc}[a+b\,x]^m\,\mathrm{Sec}[a+b\,x]^n\,dx \longrightarrow -\frac{\mathrm{Csc}[a+b\,x]^{m-1}\,\mathrm{Sec}[a+b\,x]^{n+1}}{b\,(m-1)}+\frac{n+1}{m-1}\int \mathrm{Csc}[a+b\,x]^{m-2}\,\mathrm{Sec}[a+b\,x]^{n+2}\,dx$$

- **Program code:**

```
Int[Csc[a_.+b_.*x_]^m_*Sec[a_.+b_.*x_]^n_,x_Symbol] :=
  -Csc[a+b*x]^(m-1)*Sec[a+b*x]^(n+1)/(b*(m-1)) +
  Dist[(n+1)/(m-1),Int[Csc[a+b*x]^(m-2)*Sec[a+b*x]^(n+2),x]] /;
FreeQ[{a,b},x] && RationalQ[{m,n}] && m>1 && n<-1
```

- **Reference: G&R 2.510.3, CRC 334a, A&S 4.3.128b**

- **Rule: If $m > 1 \bigwedge \frac{m+n}{2} \notin \mathbb{Z} \bigwedge \neg\left(\frac{m}{2}, \frac{n-1}{2} \in \mathbb{Z} \bigwedge n > 1\right)$, then**

$$\int \mathrm{Csc}[a+b\,x]^m\,\mathrm{Sec}[a+b\,x]^n\,dx \longrightarrow -\frac{\mathrm{Csc}[a+b\,x]^{m-1}\,\mathrm{Sec}[a+b\,x]^{n-1}}{b\,(m-1)}+\frac{m+n-2}{m-1}\int \mathrm{Csc}[a+b\,x]^{m-2}\,\mathrm{Sec}[a+b\,x]^{n}\,dx$$

- **Program code:**

```
Int[Csc[a_.+b_.*x_]^m_*Sec[a_.+b_.*x_]^n_.,x_Symbol] :=
  -Csc[a+b*x]^(m-1)*Sec[a+b*x]^(n-1)/(b*(m-1)) +
  Dist[(m+n-2)/(m-1),Int[Csc[a+b*x]^(m-2)*Sec[a+b*x]^n,x]] /;
FreeQ[{a,b,n},x] && RationalQ[m] && m>1 && Not[EvenQ[m+n]] && Not[EvenQ[m] && OddQ[n] && n>1]
```

- **Reference: G&R 2.510.2, CRC 323b, A&S 4.3.127b**

- **Rule: If $m < -1 \land m + n \neq 0$, then**

$$\int \mathrm{Csc}[a+b\,x]^m\,\mathrm{Sec}[a+b\,x]^n\,dx \longrightarrow \frac{\mathrm{Csc}[a+b\,x]^{m+1}\,\mathrm{Sec}[a+b\,x]^{n-1}}{b\,(m+n)}+\frac{m+1}{m+n}\int \mathrm{Csc}[a+b\,x]^{m+2}\,\mathrm{Sec}[a+b\,x]^{n}\,dx$$

- **Program code:**

```
Int[Csc[a_.+b_.*x_]^m_*Sec[a_.+b_.*x_]^n_.,x_Symbol] :=
  Csc[a+b*x]^(m+1)*Sec[a+b*x]^(n-1)/(b*(m+n)) +
  Dist[(m+1)/(m+n),Int[Csc[a+b*x]^(m+2)*Sec[a+b*x]^n,x]] /;
FreeQ[{a,b,n},x] && RationalQ[m] && m<-1 && NonzeroQ[m+n]
```

$$\int \mathrm{Sec}[a+b\,x]^m\,\mathrm{Tan}[a+b\,x]^n\,dx$$

- **Derivation: Power rule for integration**
- **Rule:**

$$\int \mathrm{Sec}[a+b\,x]^m\,\mathrm{Tan}[a+b\,x]\,dx \longrightarrow \frac{\mathrm{Sec}[a+b\,x]^m}{b\,m}$$

- **Program code:**

```
Int[Sec[a_.+b_.*x_]^m_.*Tan[a_.+b_.*x_]^n_.,x_Symbol] :=
  Sec[a+b*x]^m/(b*m) /;
FreeQ[{a,b,m},x] && n===1
```

```
Int[Csc[a_.+b_.*x_]^m_.*Cot[a_.+b_.*x_]^n_.,x_Symbol] :=
  -Csc[a+b*x]^m/(b*m) /;
FreeQ[{a,b,m},x] && n===1
```

- **Derivation: Integration by substitution**
- **Basis:** If $\frac{m}{2}\in\mathbb{Z}$, then $\mathrm{Sec}[z]^m = \left(1+\mathrm{Tan}[z]^2\right)^{\frac{m-2}{2}}\,\mathrm{Tan}'[z]$
- **Rule:** If $\frac{m}{2}\in\mathbb{Z}\bigwedge m>2\bigwedge \neg\left(\frac{n-1}{2}\in\mathbb{Z}\bigwedge 0<n<m-1\right)$, then

$$\int \mathrm{Sec}[a+b\,x]^m\,\mathrm{Tan}[a+b\,x]^n\,dx \longrightarrow \frac{1}{b}\,\mathrm{Subst}\left[\mathrm{Int}\left[x^n\left(1+x^2\right)^{\frac{m-2}{2}},\,x\right],\,x,\,\mathrm{Tan}[a+b\,x]\right]$$

- **Program code:**

```
Int[Sec[a_.+b_.*x_]^m_*Tan[a_.+b_.*x_]^n_.,x_Symbol] :=
  Dist[1/b,Subst[Int[Regularize[x^n*(1+x^2)^((m-2)/2),x],x],x,Tan[a+b*x]]] /;
FreeQ[{a,b,n},x] && EvenQ[m] && m>2 && Not[OddQ[n] && 0<n<m-1]
```

```
Int[Csc[a_.+b_.*x_]^m_*Cot[a_.+b_.*x_]^n_.,x_Symbol] :=
  Dist[-1/b,Subst[Int[Regularize[x^n*(1+x^2)^((m-2)/2),x],x],x,Cot[a+b*x]]] /;
FreeQ[{a,b,n},x] && EvenQ[m] && m>2 && Not[OddQ[n] && 0<n<m-1]
```

- **Derivation: Integration by substitution**

- **Basis: If** $\frac{n-1}{2} \in \mathbb{Z}$, **then** $\text{Sec}[z]^m \, \text{Tan}[z]^n = \text{Sec}[z]^{m-1} \left(-1 + \text{Sec}[z]^2\right)^{\frac{n-1}{2}} \text{Sec}'[z]$

- **Rule: If** $\frac{n-1}{2} \in \mathbb{Z} \bigwedge \neg \left(\frac{m}{2} \in \mathbb{Z} \bigwedge 0 < m \le n + 1\right)$, **then**

$$\int \text{Sec}[a + b\,x]^m \, \text{Tan}[a + b\,x]^n \, dx \longrightarrow \frac{1}{b} \, \text{Subst}\left[\int x^{m-1} \left(-1 + x^2\right)^{\frac{n-1}{2}} dx, \, x, \, \text{Sec}[a + b\,x]\right]$$

- **Program code:**

```
Int[Sec[a_.+b_.*x_]^m_.*Tan[a_.+b_.*x_]^n_.,x_Symbol] :=
  Dist[1/b,Subst[Int[Regularize[x^(m-1)*(-1+x^2)^((n-1)/2),x],x],x,Sec[a+b*x]]] /;
FreeQ[{a,b,m},x] && OddQ[n] && Not[EvenQ[m] && 0<m<=n+1]
```

```
Int[Csc[a_.+b_.*x_]^m_.*Cot[a_.+b_.*x_]^n_.,x_Symbol] :=
  Dist[-1/b,Subst[Int[Regularize[x^(m-1)*(-1+x^2)^((n-1)/2),x],x],x,Csc[a+b*x]]] /;
FreeQ[{a,b,m},x] && OddQ[n] && Not[EvenQ[m] && 0<m<=n+1]
```

- **Reference: G&R 2.510.3, CRC 334a**

- **Rule: If** $m > 1 \bigwedge n < -1 \bigwedge \frac{m}{2} \notin \mathbb{Z}$, **then**

$$\int \text{Sec}[a + b\,x]^m \, \text{Tan}[a + b\,x]^n \, dx \longrightarrow \frac{\text{Sec}[a + b\,x]^{m-2} \, \text{Tan}[a + b\,x]^{n+1}}{b\,(n + 1)} - \frac{m - 2}{n + 1} \int \text{Sec}[a + b\,x]^{m-2} \, \text{Tan}[a + b\,x]^{n+2} \, dx$$

- **Program code:**

```
Int[Sec[a_.+b_.*x_]^m_*Tan[a_.+b_.*x_]^n_,x_Symbol] :=
  Sec[a+b*x]^(m-2)*Tan[a+b*x]^(n+1)/(b*(n+1)) -
  Dist[(m-2)/(n+1),Int[Sec[a+b*x]^(m-2)*Tan[a+b*x]^(n+2),x]] /;
FreeQ[{a,b},x] && RationalQ[{m,n}] && m>1 && n<-1 && Not[EvenQ[m]]
```

- **Reference: G&R 2.510.6, CRC 334b**

```
Int[Csc[a_.+b_.*x_]^m_*Cot[a_.+b_.*x_]^n_,x_Symbol] :=
  -Csc[a+b*x]^(m-2)*Cot[a+b*x]^(n+1)/(b*(n+1)) -
  Dist[(m-2)/(n+1),Int[Csc[a+b*x]^(m-2)*Cot[a+b*x]^(n+2),x]] /;
FreeQ[{a,b},x] && RationalQ[{m,n}] && m>1 && n<-1 && Not[EvenQ[m]]
```

- **Reference: G&R 2.510.2, CRC 323b**
- **Rule: If** $m < -1 \bigwedge n > 1 \bigwedge \frac{m}{2} \notin \mathbb{Z}$**, then**

$$\int \text{Sec}[a+b\,x]^m\,\text{Tan}[a+b\,x]^n\,dx \longrightarrow \frac{\text{Sec}[a+b\,x]^m\,\text{Tan}[a+b\,x]^{n-1}}{b\,m} - \frac{n-1}{m}\int \text{Sec}[a+b\,x]^{m+2}\,\text{Tan}[a+b\,x]^{n-2}\,dx$$

- **Program code:**

```
Int[Sec[a_.+b_.*x_]^m_*Tan[a_.+b_.*x_]^n_,x_Symbol] :=
  Sec[a+b*x]^m*Tan[a+b*x]^(n-1)/(b*m) -
  Dist[(n-1)/m,Int[Sec[a+b*x]^(m+2)*Tan[a+b*x]^(n-2),x]] /;
FreeQ[{a,b},x] && RationalQ[{m,n}] && m<-1 && n>1 && Not[EvenQ[m]]
```

- **Reference: G&R 2.510.5, CRC 323a**

```
Int[Csc[a_.+b_.*x_]^m_*Cot[a_.+b_.*x_]^n_,x_Symbol] :=
  -Csc[a+b*x]^m*Cot[a+b*x]^(n-1)/(b*m) -
  Dist[(n-1)/m,Int[Csc[a+b*x]^(m+2)*Cot[a+b*x]^(n-2),x]] /;
FreeQ[{a,b},x] && RationalQ[{m,n}] && m<-1 && n>1 && Not[EvenQ[m]]
```

- **Reference: G&R 2.510.5, CRC 323a**
- **Rule: If** $m + n + 1 = 0$**, then**

$$\int \text{Sec}[a+b\,x]^m\,\text{Tan}[a+b\,x]^n\,dx \longrightarrow -\frac{\text{Sec}[a+b\,x]^m\,\text{Tan}[a+b\,x]^{n+1}}{b\,m}$$

- **Program code:**

```
Int[Sec[a_.+b_.*x_]^m_*Tan[a_.+b_.*x_]^n_,x_Symbol] :=
  -Sec[a+b*x]^m*Tan[a+b*x]^(n+1)/(b*m) /;
FreeQ[{a,b,m,n},x] && ZeroQ[m+n+1]
```

- **Reference: G&R 2.510.2, CRC 323b**

```
Int[Csc[a_.+b_.*x_]^m_.*Cot[a_.+b_.*x_]^n_,x_Symbol] :=
  Csc[a+b*x]^m*Cot[a+b*x]^(n+1)/(b*m) /;
FreeQ[{a,b,m,n},x] && ZeroQ[m+n+1]
```

- **Rule: If** $m < -1 \bigwedge \frac{m}{2} \notin \mathbb{Z}$, **then**

$$\int \operatorname{Sec}[a+b\,x]^m \operatorname{Tan}[a+b\,x]^n \,dx \longrightarrow -\frac{\operatorname{Sec}[a+b\,x]^m \operatorname{Tan}[a+b\,x]^{n+1}}{b\,m} + \frac{m+n+1}{m} \int \operatorname{Sec}[a+b\,x]^{m+2} \operatorname{Tan}[a+b\,x]^n \,dx$$

- **Program code:**

```
Int[Sec[a_.+b_.*x_]^m_*Tan[a_.+b_.*x_]^n_,x_Symbol] :=
  -Sec[a+b*x]^m*Tan[a+b*x]^(n+1)/(b*m) +
  Dist[(m+n+1)/m,Int[Sec[a+b*x]^(m+2)*Tan[a+b*x]^n,x]] /;
FreeQ[{a,b,n},x] && RationalQ[m] && m<-1 && Not[EvenQ[m]]
```

```
Int[Csc[a_.+b_.*x_]^m_*Cot[a_.+b_.*x_]^n_,x_Symbol] :=
  Csc[a+b*x]^m*Cot[a+b*x]^(n+1)/(b*m) +
  Dist[(m+n+1)/m,Int[Csc[a+b*x]^(m+2)*Cot[a+b*x]^n,x]] /;
FreeQ[{a,b,n},x] && RationalQ[m] && m<-1 && Not[EvenQ[m]]
```

- **Reference: G&R 2.510.6, CRC 334b**

- **Rule: If** $m > 1 \bigwedge m+n-1 \neq 0 \bigwedge \frac{m}{2} \notin \mathbb{Z} \bigwedge \frac{n-1}{2} \notin \mathbb{Z}$, **then**

$$\int \operatorname{Sec}[a+b\,x]^m \operatorname{Tan}[a+b\,x]^n \,dx \longrightarrow \frac{\operatorname{Sec}[a+b\,x]^{m-2} \operatorname{Tan}[a+b\,x]^{n+1}}{b\,(m+n-1)} + \frac{m-2}{m+n-1} \int \operatorname{Sec}[a+b\,x]^{m-2} \operatorname{Tan}[a+b\,x]^n \,dx$$

- **Program code:**

```
Int[Sec[a_.+b_.*x_]^m_*Tan[a_.+b_.*x_]^n_,x_Symbol] :=
  Sec[a+b*x]^(m-2)*Tan[a+b*x]^(n+1)/(b*(m+n-1)) +
  Dist[(m-2)/(m+n-1),Int[Sec[a+b*x]^(m-2)*Tan[a+b*x]^n,x]] /;
FreeQ[{a,b,n},x] && RationalQ[m] && m>1 && NonzeroQ[m+n-1] && Not[EvenQ[m]] && Not[OddQ[n]]
```

- **Reference: G&R 2.510.3, CRC 334a**

```
Int[Csc[a_.+b_.*x_]^m_*Cot[a_.+b_.*x_]^n_,x_Symbol] :=
  -Csc[a+b*x]^(m-2)*Cot[a+b*x]^(n+1)/(b*(m+n-1)) +
  Dist[(m-2)/(m+n-1),Int[Csc[a+b*x]^(m-2)*Cot[a+b*x]^n,x]] /;
FreeQ[{a,b,n},x] && RationalQ[m] && m>1 && NonzeroQ[m+n-1] && Not[EvenQ[m]] && Not[OddQ[n]]
```

- **Reference: G&R 2.510.1**
- **Rule: If** $n > 1 \bigwedge m + n - 1 \neq 0 \bigwedge \frac{m}{2} \notin \mathbb{Z} \bigwedge \frac{n-1}{2} \notin \mathbb{Z}$**, then**

$$\int \mathrm{Sec}[a + b\,x]^m\,\mathrm{Tan}[a + b\,x]^n\,dx \longrightarrow \frac{\mathrm{Sec}[a + b\,x]^m\,\mathrm{Tan}[a + b\,x]^{n-1}}{b\,(m + n - 1)} - \frac{n - 1}{m + n - 1}\int \mathrm{Sec}[a + b\,x]^m\,\mathrm{Tan}[a + b\,x]^{n-2}\,dx$$

- **Program code:**

```
Int[Sec[a_.+b_.*x_]^m_.*Tan[a_.+b_.*x_]^n_,x_Symbol] :=
  Sec[a+b*x]^m*Tan[a+b*x]^(n-1)/(b*(m+n-1)) -
  Dist[(n-1)/(m+n-1),Int[Sec[a+b*x]^m*Tan[a+b*x]^(n-2),x]] /;
FreeQ[{a,b,m},x] && RationalQ[n] && n>1 && NonzeroQ[m+n-1] && Not[EvenQ[m]] && Not[OddQ[n]]
```

- **Reference: G&R 2.510.4**

```
Int[Csc[a_.+b_.*x_]^m_.*Cot[a_.+b_.*x_]^n_,x_Symbol] :=
  -Csc[a+b*x]^m*Cot[a+b*x]^(n-1)/(b*(m+n-1)) -
  Dist[(n-1)/(m+n-1),Int[Csc[a+b*x]^m*Cot[a+b*x]^(n-2),x]] /;
FreeQ[{a,b,m},x] && RationalQ[n] && n>1 && NonzeroQ[m+n-1] && Not[EvenQ[m]] && Not[OddQ[n]]
```

- **Reference: G&R 2.510.4**
- **Rule: If** $n < -1 \bigwedge \frac{m}{2} \notin \mathbb{Z}$**, then**

$$\int \mathrm{Sec}[a + b\,x]^m\,\mathrm{Tan}[a + b\,x]^n\,dx \longrightarrow \frac{\mathrm{Sec}[a + b\,x]^m\,\mathrm{Tan}[a + b\,x]^{n+1}}{b\,(n + 1)} - \frac{m + n + 1}{n + 1}\int \mathrm{Sec}[a + b\,x]^m\,\mathrm{Tan}[a + b\,x]^{n+2}\,dx$$

- **Program code:**

```
Int[Sec[a_.+b_.*x_]^m_*Tan[a_.+b_.*x_]^n_,x_Symbol] :=
  Sec[a+b*x]^m*Tan[a+b*x]^(n+1)/(b*(n+1)) -
  Dist[(m+n+1)/(n+1),Int[Sec[a+b*x]^m*Tan[a+b*x]^(n+2),x]] /;
FreeQ[{a,b,m},x] && RationalQ[n] && n<-1 && Not[EvenQ[m]]
```

- **Reference: G&R 2.510.1**

```
Int[Csc[a_.+b_.*x_]^m_.*Cot[a_.+b_.*x_]^n_,x_Symbol] :=
  -Csc[a+b*x]^m*Cot[a+b*x]^(n+1)/(b*(n+1)) -
  Dist[(m+n+1)/(n+1),Int[Csc[a+b*x]^m*Cot[a+b*x]^(n+2),x]] /;
FreeQ[{a,b,m},x] && RationalQ[n] && n<-1 && Not[EvenQ[m]]
```

$$\int x^m \operatorname{Sec}[a+b\,x^n]^p \operatorname{Sin}[a+b\,x^n]\,dx$$

- **Derivation: Integration by parts**
- **Rule: If** $n \in \mathbb{Z} \land m-n \geq 0 \land p-1 \neq 0$**, then**

$$\int x^m \operatorname{Sec}[a+b\,x^n]^p \operatorname{Sin}[a+b\,x^n]\,dx \longrightarrow \frac{x^{m-n+1} \operatorname{Sec}[a+b\,x^n]^{p-1}}{b\,n\,(p-1)} - \frac{m-n+1}{b\,n\,(p-1)} \int x^{m-n} \operatorname{Sec}[a+b\,x^n]^{p-1}\,dx$$

- **Program code:**

```
Int[x_^m_.*Sec[a_.+b_.*x_^n_.]^p_*Sin[a_.+b_.*x_^n_.],x_Symbol] :=
  x^(m-n+1)*Sec[a+b*x^n]^(p-1)/(b*n*(p-1)) -
  Dist[(m-n+1)/(b*n*(p-1)),Int[x^(m-n)*Sec[a+b*x^n]^(p-1),x]] /;
FreeQ[{a,b,p},x] && RationalQ[m] && IntegerQ[n] && m-n>=0 && NonzeroQ[p-1]
```

```
Int[x_^m_.*Csc[a_.+b_.*x_^n_.]^p_*Cos[a_.+b_.*x_^n_.],x_Symbol] :=
  -x^(m-n+1)*Csc[a+b*x^n]^(p-1)/(b*n*(p-1)) +
  Dist[(m-n+1)/(b*n*(p-1)),Int[x^(m-n)*Csc[a+b*x^n]^(p-1),x]] /;
FreeQ[{a,b,p},x] && RationalQ[m] && IntegerQ[n] && m-n>=0 && NonzeroQ[p-1]
```

$$\int x^m \operatorname{Sec}[a + b x^n]^p \operatorname{Tan}[a + b x^n] \, dx$$

- **Derivation: Integration by parts**
- **Note: Dummy exponent $q = 1$ required in program code so InputForm of integrand is recognized.**
- **Rule: If $n \in \mathbb{Z} \land m - n \geq 0$, then**

$$\int x^m \operatorname{Sec}[a + b x^n]^p \operatorname{Tan}[a + b x^n] \, dx \longrightarrow \frac{x^{m-n+1} \operatorname{Sec}[a + b x^n]^p}{b n p} - \frac{m - n + 1}{b n p} \int x^{m-n} \operatorname{Sec}[a + b x^n]^p \, dx$$

- **Program code:**

```
Int[x_^m_.*Sec[a_.+b_.*x_^n_.]^p_.*Tan[a_.+b_.*x_^n_.]^q_.,x_Symbol] :=
  x^(m-n+1)*Sec[a+b*x^n]^p/(b*n*p) -
  Dist[(m-n+1)/(b*n*p),Int[x^(m-n)*Sec[a+b*x^n]^p,x]] /;
FreeQ[{a,b,p},x] && RationalQ[m] && IntegerQ[n] && m-n>=0 && q===1
```

```
Int[x_^m_.*Csc[a_.+b_.*x_^n_.]^p_.*Cot[a_.+b_.*x_^n_.]^q_.,x_Symbol] :=
  -x^(m-n+1)*Csc[a+b*x^n]^p/(b*n*p) +
  Dist[(m-n+1)/(b*n*p),Int[x^(m-n)*Csc[a+b*x^n]^p,x]] /;
FreeQ[{a,b,p},x] && RationalQ[m] && IntegerQ[n] && m-n>=0 && q===1
```

$$\int \mathrm{Sec}[a+b\,\mathrm{Log}[c\,x^n]]^p\,dx$$

- **Rule: If** $p-1\neq 0\land b^2 n^2 (p-2)^2+1=0$**, then**

$$\int \mathrm{Sec}[a+b\,\mathrm{Log}[c\,x^n]]^p\,dx\;\longrightarrow\;\frac{x\,(b\,n\,(p-2)+\mathrm{Tan}[a+b\,\mathrm{Log}[c\,x^n]])\,\mathrm{Sec}[a+b\,\mathrm{Log}[c\,x^n]]^{p-2}}{b\,n\,(p-1)}$$

- **Program code:**

```
Int[Sec[a_.+b_.*Log[c_.*x_^n_.]]^p_,x_Symbol] :=
  x*(b*n*(p-2)+Tan[a+b*Log[c*x^n]])*Sec[a+b*Log[c*x^n]]^(p-2)/(b*n*(p-1)) /;
FreeQ[{a,b,c,n,p},x] && NonzeroQ[p-1] && ZeroQ[b^2*n^2*(p-2)^2+1]
```

```
Int[Csc[a_.+b_.*Log[c_.*x_^n_.]]^p_,x_Symbol] :=
  x*(b*n*(p-2)-Cot[a+b*Log[c*x^n]])*Csc[a+b*Log[c*x^n]]^(p-2)/(b*n*(p-1)) /;
FreeQ[{a,b,c,n,p},x] && NonzeroQ[p-1] && ZeroQ[b^2*n^2*(p-2)^2+1]
```

- **Rule: If** $p>1\land p\neq 2\land b^2 n^2 (p-2)^2+1\neq 0$**, then**

$$\int \mathrm{Sec}[a+b\,\mathrm{Log}[c\,x^n]]^p\,dx\;\longrightarrow\;\frac{x\,\mathrm{Tan}[a+b\,\mathrm{Log}[c\,x^n]]\,\mathrm{Sec}[a+b\,\mathrm{Log}[c\,x^n]]^{p-2}}{b\,n\,(p-1)}-\frac{x\,\mathrm{Sec}[a+b\,\mathrm{Log}[c\,x^n]]^{p-2}}{b^2 n^2 (p-1)(p-2)}+\frac{b^2 n^2 (p-2)^2+1}{b^2 n^2 (p-1)(p-2)}\int \mathrm{Sec}[a+b\,\mathrm{Log}[c\,x^n]]^{p-2}\,dx$$

- **Program code:**

```
Int[Sec[a_.+b_.*Log[c_.*x_^n_.]]^p_,x_Symbol] :=
  x*Tan[a+b*Log[c*x^n]]*Sec[a+b*Log[c*x^n]]^(p-2)/(b*n*(p-1)) -
  x*Sec[a+b*Log[c*x^n]]^(p-2)/(b^2*n^2*(p-1)*(p-2)) +
  Dist[(b^2*n^2*(p-2)^2+1)/(b^2*n^2*(p-1)*(p-2)),Int[Sec[a+b*Log[c*x^n]]^(p-2),x]] /;
FreeQ[{a,b,c,n},x] && RationalQ[p] && p>1 && p≠2 && NonzeroQ[b^2*n^2*(p-2)^2+1]
```

```
Int[Csc[a_.+b_.*Log[c_.*x_^n_.]]^p_,x_Symbol] :=
  -x*Cot[a+b*Log[c*x^n]]*Csc[a+b*Log[c*x^n]]^(p-2)/(b*n*(p-1)) -
  x*Csc[a+b*Log[c*x^n]]^(p-2)/(b^2*n^2*(p-1)*(p-2)) +
  Dist[(b^2*n^2*(p-2)^2+1)/(b^2*n^2*(p-1)*(p-2)),Int[Csc[a+b*Log[c*x^n]]^(p-2),x]] /;
FreeQ[{a,b,c,n},x] && RationalQ[p] && p>1 && p≠2 && NonzeroQ[b^2*n^2*(p-2)^2+1]
```

- **Rule: If** $p < -1 \wedge b^2 n^2 p^2 + 1 \neq 0$, **then**

$$\int \mathrm{Sec}[a + b\,\mathrm{Log}[c\,x^n]]^p\,dx \longrightarrow -\frac{b\,n\,p\,x\,\mathrm{Sin}[a + b\,\mathrm{Log}[c\,x^n]]\,\mathrm{Sec}[a + b\,\mathrm{Log}[c\,x^n]]^{p+1}}{b^2 n^2 p^2 + 1} + \frac{x\,\mathrm{Sec}[a + b\,\mathrm{Log}[c\,x^n]]^p}{b^2 n^2 p^2 + 1} + \frac{b^2 n^2 p\,(p+1)}{b^2 n^2 p^2 + 1}\int \mathrm{Sec}[a + b\,\mathrm{Log}[c\,x^n]]^{p+2}\,dx$$

- **Program code:**

```
Int[Sec[a_.+b_.*Log[c_.*x_^n_.]]^p_,x_Symbol] :=
  -b*n*p*x*Sin[a+b*Log[c*x^n]]*Sec[a+b*Log[c*x^n]]^(p+1)/(b^2*n^2*p^2+1) +
  x*Sec[a+b*Log[c*x^n]]^p/(b^2*n^2*p^2+1) +
  Dist[b^2*n^2*p*(p+1)/(b^2*n^2*p^2+1),Int[Sec[a+b*Log[c*x^n]]^(p+2),x]] /;
FreeQ[{a,b,c,n},x] && RationalQ[p] && p<-1 && NonzeroQ[b^2*n^2*p^2+1]
```

```
Int[Csc[a_.+b_.*Log[c_.*x_^n_.]]^p_,x_Symbol] :=
  b*n*p*x*Cos[a+b*Log[c*x^n]]*Csc[a+b*Log[c*x^n]]^(p+1)/(b^2*n^2*p^2+1) +
  x*Csc[a+b*Log[c*x^n]]^p/(b^2*n^2*p^2+1) +
  Dist[b^2*n^2*p*(p+1)/(b^2*n^2*p^2+1),Int[Csc[a+b*Log[c*x^n]]^(p+2),x]] /;
FreeQ[{a,b,c,n},x] && RationalQ[p] && p<-1 && NonzeroQ[b^2*n^2*p^2+1]
```

$$\int x^m \operatorname{Sec}[a+b\operatorname{Log}[c\,x^n]]^p\,dx$$

- **Rule: If** $m+1\neq 0\land p-1\neq 0\land b^2 n^2 (p-2)^2+(m+1)^2=0$**, then**

$$\int x^m \operatorname{Sec}[a+b\operatorname{Log}[c\,x^n]]^p\,dx \longrightarrow \frac{x^{m+1}\,(b\,n\,(p-2)+(m+1)\operatorname{Tan}[a+b\operatorname{Log}[c\,x^n]])\operatorname{Sec}[a+b\operatorname{Log}[c\,x^n]]^{p-2}}{b\,n\,(m+1)\,(p-1)}$$

- **Program code:**

```
Int[x_^m_.*Sec[a_.+b_.*Log[c_.*x_^n_.]]^p_,x_Symbol] :=
  x^(m+1)*(b*n*(p-2)+(m+1)*Tan[a+b*Log[c*x^n]])*Sec[a+b*Log[c*x^n]]^(p-2)/(b*n*(m+1)*(p-1)) /;
FreeQ[{a,b,c,m,n,p},x] && NonzeroQ[m+1] && NonzeroQ[p-1] && ZeroQ[b^2*n^2*(p-2)^2+(m+1)^2]
```

```
Int[x_^m_.*Csc[a_.+b_.*Log[c_.*x_^n_.]]^p_,x_Symbol] :=
  x^(m+1)*(b*n*(p-2)-(m+1)*Cot[a+b*Log[c*x^n]])*Csc[a+b*Log[c*x^n]]^(p-2)/(b*n*(m+1)*(p-1)) /;
FreeQ[{a,b,c,m,n,p},x] && NonzeroQ[m+1] && NonzeroQ[p-1] && ZeroQ[b^2*n^2*(p-2)^2+(m+1)^2]
```

- **Rule: If** $p>1\land p\neq 2\land b^2 n^2 (p-2)^2+(m+1)^2\neq 0$**, then**

$$\int x^m \operatorname{Sec}[a+b\operatorname{Log}[c\,x^n]]^p\,dx \longrightarrow \frac{x^{m+1}\operatorname{Tan}[a+b\operatorname{Log}[c\,x^n]]\operatorname{Sec}[a+b\operatorname{Log}[c\,x^n]]^{p-2}}{b\,n\,(p-1)} - \frac{(m+1)\,x^{m+1}\operatorname{Sec}[a+b\operatorname{Log}[c\,x^n]]^{p-2}}{b^2 n^2\,(p-1)\,(p-2)} + \frac{b^2 n^2\,(p-2)^2+(m+1)^2}{b^2 n^2\,(p-1)\,(p-2)}\int x^m \operatorname{Sec}[a+b\operatorname{Log}[c\,x^n]]^{p-2}\,dx$$

- **Program code:**

```
Int[x_^m_.*Sec[a_.+b_.*Log[c_.*x_^n_.]]^p_,x_Symbol] :=
  x^(m+1)*Tan[a+b*Log[c*x^n]]*Sec[a+b*Log[c*x^n]]^(p-2)/(b*n*(p-1)) -
  (m+1)*x^(m+1)*Sec[a+b*Log[c*x^n]]^(p-2)/(b^2*n^2*(p-1)*(p-2)) +
  Dist[(b^2*n^2*(p-2)^2+(m+1)^2)/(b^2*n^2*(p-1)*(p-2)),Int[x^m*Sec[a+b*Log[c*x^n]]^(p-2),x]] /;
FreeQ[{a,b,c,m,n},x] && RationalQ[p] && p>1 && p≠2 && NonzeroQ[b^2*n^2*(p-2)^2+(m+1)^2]
```

```
Int[x_^m_.*Csc[a_.+b_.*Log[c_.*x_^n_.]]^p_,x_Symbol] :=
  -x^(m+1)*Cot[a+b*Log[c*x^n]]*Csc[a+b*Log[c*x^n]]^(p-2)/(b*n*(p-1)) -
  (m+1)*x^(m+1)*Csc[a+b*Log[c*x^n]]^(p-2)/(b^2*n^2*(p-1)*(p-2)) +
  Dist[(b^2*n^2*(p-2)^2+(m+1)^2)/(b^2*n^2*(p-1)*(p-2)),Int[x^m*Csc[a+b*Log[c*x^n]]^(p-2),x]] /;
FreeQ[{a,b,c,m,n},x] && RationalQ[p] && p>1 && p≠2 && NonzeroQ[b^2*n^2*(p-2)^2+(m+1)^2]
```

- **Rule: If** $p<-1 \wedge b^2 n^2 p^2+(m+1)^2 \neq 0$, **then**

$$\int x^m \operatorname{Sec}[a+b \operatorname{Log}[c x^n]]^p \, dx \longrightarrow -\frac{b\,n\,p\,x^{m+1} \operatorname{Sin}[a+b \operatorname{Log}[c x^n]] \operatorname{Sec}[a+b \operatorname{Log}[c x^n]]^{p+1}}{b^2 n^2 p^2+(m+1)^2}+\frac{(m+1)\,x^{m+1} \operatorname{Sec}[a+b \operatorname{Log}[c x^n]]^p}{b^2 n^2 p^2+(m+1)^2}+\frac{b^2 n^2 p\,(p+1)}{b^2 n^2 p^2+(m+1)^2} \int x^m \operatorname{Sec}[a+b \operatorname{Log}[c x^n]]^{p+2} \, dx$$

- **Program code:**

```
Int[x_^m_.*Sec[a_.+b_.*Log[c_.*x_^n_.]]^p_,x_Symbol] :=
  -b*n*p*x^(m+1)*Sin[a+b*Log[c*x^n]]*Sec[a+b*Log[c*x^n]]^(p+1)/(b^2*n^2*p^2+(m+1)^2) +
  (m+1)*x^(m+1)*Sec[a+b*Log[c*x^n]]^p/(b^2*n^2*p^2+(m+1)^2) +
  Dist[b^2*n^2*p*(p+1)/(b^2*n^2*p^2+(m+1)^2),Int[x^m*Sec[a+b*Log[c*x^n]]^(p+2),x]] /;
FreeQ[{a,b,c,m,n},x] && RationalQ[p] && p<-1 && NonzeroQ[b^2*n^2*p^2+(m+1)^2]
```

```
Int[x_^m_.*Csc[a_.+b_.*Log[c_.*x_^n_.]]^p_,x_Symbol] :=
  b*n*p*x^(m+1)*Cos[a+b*Log[c*x^n]]*Csc[a+b*Log[c*x^n]]^(p+1)/(b^2*n^2*p^2+(m+1)^2) +
  (m+1)*x^(m+1)*Csc[a+b*Log[c*x^n]]^p/(b^2*n^2*p^2+(m+1)^2) +
  Dist[b^2*n^2*p*(p+1)/(b^2*n^2*p^2+(m+1)^2),Int[x^m*Csc[a+b*Log[c*x^n]]^(p+2),x]] /;
FreeQ[{a,b,c,m,n},x] && RationalQ[p] && p<-1 && NonzeroQ[b^2*n^2*p^2+(m+1)^2]
```

$$\int u\,\mathrm{Csc}[a+b\,x]^n\,dx$$

- **Derivation: Algebraic simplification**
- **Basis:** $\mathrm{Csc}[2\,z] = \frac{1}{2}\,\mathrm{Csc}[z]\,\mathrm{Sec}[z]$
- **Rule: If** $n \in \mathbb{Z}$ **and** u **is a function of trig functions of** $\frac{a}{2}+\frac{b\,x}{2}$**, then**

$$\int u\,\mathrm{Csc}[a+b\,x]^n\,dx \longrightarrow \frac{1}{2^n}\int u\,\mathrm{Csc}\left[\frac{a}{2}+\frac{b\,x}{2}\right]^n\mathrm{Sec}\left[\frac{a}{2}+\frac{b\,x}{2}\right]^n dx$$

- **Program code:**

```
Int[u_*Csc[a_.+b_.*x_]^n_.,x_Symbol] :=
  Dist[1/2^n,Int[u*Csc[a/2+b*x/2]^n*Sec[a/2+b*x/2]^n,x]] /;
FreeQ[{a,b},x] && IntegerQ[n] && ZeroQ[a/2+b*x/2-FunctionOfTrig[u,x]]
```

- **Derivation: Algebraic simplification and piecewise constant extraction**
- **Basis:** $\mathrm{Csc}[2\,z] = \frac{1}{2}\,\mathrm{Csc}[z]\,\mathrm{Sec}[z]$
- **Basis:** $\partial_x \frac{\mathrm{Csc}[a+b\,x]^n}{\mathrm{Csc}\left[\frac{a}{2}+\frac{b\,x}{2}\right]^n\mathrm{Sec}\left[\frac{a}{2}+\frac{b\,x}{2}\right]^n} = 0$
- **Rule: If** $n \in \mathbb{F}$ **and** u **is a function of trig functions of** $\frac{a}{2}+\frac{b\,x}{2}$**, then**

$$\int u\,\mathrm{Csc}[a+b\,x]^n\,dx \longrightarrow \frac{\mathrm{Csc}[a+b\,x]^n}{\mathrm{Csc}\left[\frac{a}{2}+\frac{b\,x}{2}\right]^n\mathrm{Sec}\left[\frac{a}{2}+\frac{b\,x}{2}\right]^n}\int u\,\mathrm{Csc}\left[\frac{a}{2}+\frac{b\,x}{2}\right]^n\mathrm{Sec}\left[\frac{a}{2}+\frac{b\,x}{2}\right]^n dx$$

- **Program code:**

```
(* Int[u_*Csc[a_.+b_.*x_]^n_,x_Symbol] :=
  Csc[a+b*x]^n/(Csc[a/2+b*x/2]^n*Sec[a/2+b*x/2]^n)*Int[u*Csc[a/2+b*x/2]^n*Sec[a/2+b*x/2]^n,x] /;
FreeQ[{a,b},x] && FractionQ[n] && ZeroQ[a/2+b*x/2-FunctionOfTrig[u,x]] *)
```