$$\int \mathrm{Gamma}[n,\ a+b\,x]\,dx$$

- **Derivation: Integration by parts**
- **Rule:**

$$\int \mathrm{Gamma}[n,\ a+b\,x]\,dx \longrightarrow \frac{(a+b\,x)\ \mathrm{Gamma}[n,\ a+b\,x]}{b}-\frac{\mathrm{Gamma}[n+1,\ a+b\,x]}{b}$$

- **Program code:**

```
Int[Gamma[n_,a_.+b_.*x_],x_Symbol] :=
  (a+b*x)*Gamma[n,a+b*x]/b -
  Gamma[n+1,a+b*x]/b /;
FreeQ[{a,b},x]
```

$$\int x^m \text{Gamma}[n, a + b x] \, dx$$

- **Derivation: Integration by parts**
- **Rule: If** $m \in \mathbb{Z} \lor a > 0$**, then**

$$\int x^m \text{Gamma}[n, a x] \, dx \longrightarrow \frac{x^{m+1} \text{Gamma}[n, a x]}{m+1} - \frac{\text{Gamma}[m+n+1, a x]}{(m+1) \, a^{m+1}}$$

- **Program code:**

```
Int[x_^m_.*Gamma[n_,a_.*x_],x_Symbol] :=
  x^(m+1)*Gamma[n,a*x]/(m+1) -
  Gamma[m+n+1,a*x]/((m+1)*a^(m+1)) /;
FreeQ[{a,n},x] && (IntegerQ[m] || PositiveQ[a])
```

- **Derivation: Integration by parts**
- **Rule:**

$$\int x^m \text{Gamma}[n, a x] \, dx \longrightarrow \frac{x^{m+1} \text{Gamma}[n, a x]}{m+1} - \frac{x^{m+1} \text{Gamma}[m+n+1, a x]}{(m+1) \, (a x)^{m+1}}$$

- **Program code:**

```
Int[x_^m_.*Gamma[n_,a_*x_],x_Symbol] :=
  x^(m+1)*Gamma[n,a*x]/(m+1) -
  x^(m+1)*Gamma[m+n+1,a*x]/((m+1)*(a*x)^(m+1)) /;
FreeQ[{a,m,n},x]
```

- **Derivation: Integration by parts**
- **Rule: If** $m > 0$**, then**

$$\int x^m \text{Gamma}[n, a + b x] \, dx \longrightarrow \frac{x^m \, (a + b x) \, \text{Gamma}[n, a + b x]}{b \, (m+1)} - \frac{x^m \, \text{Gamma}[n+1, a + b x]}{b \, (m+1)} - \frac{a m}{b \, (m+1)} \int x^{m-1} \text{Gamma}[n, a + b x] \, dx + \frac{m}{b \, (m+1)} \int x^{m-1} \text{Gamma}[n+1, a + b x] \, dx$$

- **Program code:**

```
Int[x_^m_.*Gamma[n_,a_+b_.*x_],x_Symbol] :=
  x^m*(a+b*x)*Gamma[n,a+b*x]/(b*(m+1)) -
  x^m*Gamma[n+1,a+b*x]/(b*(m+1)) -
  Dist[a*m/(b*(m+1)),Int[x^(m-1)*Gamma[n,a+b*x],x]] +
  Dist[m/(b*(m+1)),Int[x^(m-1)*Gamma[n+1,a+b*x],x]] /;
FreeQ[{a,b,n},x] && RationalQ[m] && m>0
```

$$\int \text{LogGamma}[a + b\,x]\,dx$$

- **Derivation: Primitive rule**
- **Basis:** $\frac{\partial \psi^{(-2)}(z)}{\partial z} = \log\Gamma(z)$
- **Rule:**

$$\int \text{LogGamma}[a + b\,x]\,dx \longrightarrow \frac{\text{PolyGamma}[-2,\, a + b\,x]}{b}$$

- **Program code:**

```
Int[LogGamma[a_.+b_.*x_],x_Symbol] :=
  PolyGamma[-2,a+b*x]/b /;
FreeQ[{a,b},x]
```

$\int x^m \, \text{LogGamma}[a + b\,x] \, dx$

- **Derivation: Integration by parts**

- **Rule: If $m > 0$, then**

$$\int x^m \,\text{LogGamma}[a + b\,x]\, dx \longrightarrow \frac{x^m \,\text{PolyGamma}[-2, a + b\,x]}{b} - \frac{m}{b} \int x^{m-1} \,\text{PolyGamma}[-2, a + b\,x]\, dx$$

- **Program code:**

```
Int[x_^m_.*LogGamma[a_.+b_.*x_],x_Symbol] :=
  x^m*PolyGamma[-2,a+b*x]/b -
  Dist[m/b,Int[x^(m-1)*PolyGamma[-2,a+b*x],x]] /;
FreeQ[{a,b},x] && RationalQ[m] && m>0
```

$$\int \text{PolyGamma}[n, a + b\,x]\, dx$$

- **Derivation: Primitive rule**
- **Basis:** $\frac{\partial \psi^{(n)}(z)}{\partial z} = \psi^{(n+1)}(z)$
- **Rule:**

$$\int \text{PolyGamma}[n, a + b\,x]\, dx \longrightarrow \frac{\text{PolyGamma}[n - 1, a + b\,x]}{b}$$

- **Program code:**

```
Int[PolyGamma[n_,a_.+b_.*x_],x_Symbol] :=
  PolyGamma[n-1,a+b*x]/b /;
FreeQ[{a,b,n},x]
```

$$\int x^m \, \mathrm{PolyGamma}[n, a + b\,x] \, dx$$

- **Derivation: Integration by parts**
- **Rule: If $m > 0$, then**

$$\int x^m \, \mathrm{PolyGamma}[n, a + b\,x] \, dx \longrightarrow \frac{x^m \, \mathrm{PolyGamma}[n-1, a + b\,x]}{b} - \frac{m}{b} \int x^{m-1} \, \mathrm{PolyGamma}[n-1, a + b\,x] \, dx$$

- **Program code:**

```
Int[x_^m_.*PolyGamma[n_,a_.+b_.*x_],x_Symbol] :=
  x^m*PolyGamma[n-1,a+b*x]/b -
  Dist[m/b,Int[x^(m-1)*PolyGamma[n-1,a+b*x],x]] /;
FreeQ[{a,b,n},x] && RationalQ[m] && m>0
```

- **Derivation: Inverted integration by parts**
- **Rule: If $m < -1$, then**

$$\int x^m \, \mathrm{PolyGamma}[n, a + b\,x] \, dx \longrightarrow \frac{x^{m+1} \, \mathrm{PolyGamma}[n, a + b\,x]}{m+1} - \frac{b}{m+1} \int x^{m+1} \, \mathrm{PolyGamma}[n+1, a + b\,x] \, dx$$

- **Program code:**

```
Int[x_^m_.*PolyGamma[n_,a_.+b_.*x_],x_Symbol] :=
  x^(m+1)*PolyGamma[n,a+b*x]/(m+1) -
  Dist[b/(m+1),Int[x^(m+1)*PolyGamma[n+1,a+b*x],x]] /;
FreeQ[{a,b,n},x] && RationalQ[m] && m<-1
```

$$\int \mathrm{Gamma}[a+b\,x]^n\,\mathrm{PolyGamma}[0,\,a+b\,x]\,dx$$

- **Derivation: Primitive rule**
- **Basis:** $\frac{\partial \Gamma(z)^n}{\partial z} = n\,\psi^{(0)}(z)\,\Gamma(z)^n$
- **Rule:**

$$\int \mathrm{Gamma}[a+b\,x]^n\,\mathrm{PolyGamma}[0,\,a+b\,x]\,dx \longrightarrow \frac{\mathrm{Gamma}[a+b\,x]^n}{b\,n}$$

- **Program code:**

```
Int[Gamma[a_.+b_.*x_]^n_.*PolyGamma[0,a_.+b_.*x_],x_Symbol] :=
  Gamma[a+b*x]^n/(b*n) /;
FreeQ[{a,b,n},x]
```

- **Derivation: Primitive rule**
- **Basis:** $\frac{\partial (z!)^n}{\partial z} = n\,\psi^{(0)}(z+1)\,(z!)^n$
- **Rule:**

$$\int ((a+b\,x)!)^n\,\mathrm{PolyGamma}[0,\,1+a+b\,x]\,dx \longrightarrow \frac{((a+b\,x)!)^n}{b\,n}$$

- **Program code:**

```
Int[((a_.+b_.*x_)!)^n_.*PolyGamma[0,c_.+b_.*x_],x_Symbol] :=
  ((a+b*x)!)^n/(b*n) /;
FreeQ[{a,b,c,n},x] && ZeroQ[a-c+1]
```