$$\int (a \operatorname{Cosh}[c+d x]+b \operatorname{Sinh}[c+d x])^n \, dx$$

- **Rule: If $a^2-b^2=0$, then**

$$\int (a \operatorname{Cosh}[c+d x]+b \operatorname{Sinh}[c+d x])^n \, dx \longrightarrow \frac{a\,(a \operatorname{Cosh}[c+d x]+b \operatorname{Sinh}[c+d x])^n}{b\, d\, n}$$

- **Program code:**

```
Int[(a_.*Cosh[c_.+d_.*x_]+b_.*Sinh[c_.+d_.*x_])^n_,x_Symbol] :=
  a*(a*Cosh[c+d*x]+b*Sinh[c+d*x])^n/(b*d*n) /;
FreeQ[{a,b,c,d,n},x] && ZeroQ[a^2-b^2]
```

- **Rule: If $a^2-b^2\neq 0$, then**

$$\int \frac{1}{(a \operatorname{Cosh}[c+d x]+b \operatorname{Sinh}[c+d x])^2} \, dx \longrightarrow \frac{\operatorname{Sinh}[c+d x]}{a\, d\, (a \operatorname{Cosh}[c+d x]+b \operatorname{Sinh}[c+d x])}$$

- **Program code:**

```
Int[1/(a_.*Cosh[c_.+d_.*x_]+b_.*Sinh[c_.+d_.*x_])^2,x_Symbol] :=
  Sinh[c+d*x]/(a*d*(a*Cosh[c+d*x]+b*Sinh[c+d*x])) /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2-b^2]
```

- **Reference: G&R 2.449'**
- **Derivation: Integration by substitution**
- **Basis: If $\frac{n-1}{2}\in\mathbb{Z}$, then** $(a \operatorname{Cosh}[z]+b \operatorname{Sinh}[z])^n = \left(a^2-b^2+(b \operatorname{Cosh}[z]+a \operatorname{Sinh}[z])^2\right)^{\frac{n-1}{2}} \partial_z (b \operatorname{Cosh}[z]+a \operatorname{Sinh}[z])$
- **Note: For odd $n<-1$, might as well stay in the hyperbolic world using 2nd rule below. (???)**
- **Rule: If $a^2-b^2\neq 0 \bigwedge \frac{n-1}{2}\in\mathbb{Z} \bigwedge n>0$, then**

$$\int (a \operatorname{Cosh}[c+d x]+b \operatorname{Sinh}[c+d x])^n \, dx \longrightarrow$$
$$\frac{1}{d} \operatorname{Subst}\left[\operatorname{Int}\left[\left(a^2-b^2+x^2\right)^{\frac{n-1}{2}}, x\right], x, b \operatorname{Cosh}[c+d x]+a \operatorname{Sinh}[c+d x]\right]$$

- **Program code:**

```
Int[(a_.*Cosh[c_.+d_.*x_]+b_.*Sinh[c_.+d_.*x_])^n_,x_Symbol] :=
  Dist[1/d,Subst[Int[Regularize[(a^2-b^2+x^2)^((n-1)/2),x],x],x,b*Cosh[c+d*x]+a*Sinh[c+d*x]]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2-b^2] && OddQ[n] && n>0
```

- **Derivation: Integration by parts with a double-back flip**
- **Rule: If** $a^2 - b^2 \neq 0 \bigwedge n > 1 \bigwedge \frac{n-1}{2} \notin \mathbb{Z}$**, then**

$$\int (a \operatorname{Cosh}[c + d x] + b \operatorname{Sinh}[c + d x])^n \, dx \longrightarrow$$
$$\frac{(b \operatorname{Cosh}[c + d x] + a \operatorname{Sinh}[c + d x]) (a \operatorname{Cosh}[c + d x] + b \operatorname{Sinh}[c + d x])^{n-1}}{d n} +$$
$$\frac{(n-1) \left(a^2 - b^2\right)}{n} \int (a \operatorname{Cosh}[c + d x] + b \operatorname{Sinh}[c + d x])^{n-2} \, dx$$

- **Program code:**

```
Int[(a_.*Cosh[c_.+d_.*x_]+b_.*Sinh[c_.+d_.*x_])^n_,x_Symbol] :=
  (b*Cosh[c+d*x]+a*Sinh[c+d*x])*(a*Cosh[c+d*x]+b*Sinh[c+d*x])^(n-1)/(d*n) +
  Dist[(n-1)*(a^2-b^2)/n,Int[(a*Cosh[c+d*x]+b*Sinh[c+d*x])^(n-2),x]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2-b^2] && RationalQ[n] && n>1 && Not[OddQ[n]]
```

- **Derivation: Integration by parts with a double-back flip**
- **Rule: If** $a^2 - b^2 \neq 0 \wedge n < -1 \wedge n \neq -2$**, then**

$$\int (a \operatorname{Cosh}[c + d x] + b \operatorname{Sinh}[c + d x])^n \, dx \longrightarrow$$
$$-\frac{(b \operatorname{Cosh}[c + d x] + a \operatorname{Sinh}[c + d x]) (a \operatorname{Cosh}[c + d x] + b \operatorname{Sinh}[c + d x])^{n+1}}{d (n+1) \left(a^2 - b^2\right)} +$$
$$\frac{n+2}{(n+1) \left(a^2 - b^2\right)} \int (a \operatorname{Cosh}[c + d x] + b \operatorname{Sinh}[c + d x])^{n+2} \, dx$$

- **Program code:**

```
Int[(a_.*Cosh[c_.+d_.*x_]+b_.*Sinh[c_.+d_.*x_])^n_,x_Symbol] :=
  -(b*Cosh[c+d*x]+a*Sinh[c+d*x])*(a*Cosh[c+d*x]+b*Sinh[c+d*x])^(n+1)/(d*(n+1)*(a^2-b^2)) +
  Dist[(n+2)/((n+1)*(a^2-b^2)),Int[(a*Cosh[c+d*x]+b*Sinh[c+d*x])^(n+2),x]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2-b^2] && RationalQ[n] && n<-1 && n≠-2
```

$$\int \frac{\text{Cosh}[c+d\,x]^m\,\text{Sinh}[c+d\,x]^n}{(a\,\text{Cosh}[c+d\,x]+b\,\text{Sinh}[c+d\,x])}\,dx$$

- **Derivation: Algebraic expansion**

- **Basis:** $\frac{\text{Cosh}[z]\,\text{Sinh}[z]}{a\,\text{Cosh}[z]+b\,\text{Sinh}[z]} = -\frac{b\,\text{Cosh}[z]}{a^2-b^2} + \frac{a\,\text{Sinh}[z]}{a^2-b^2} + \frac{a\,b}{(a^2-b^2)\,(a\,\text{Cosh}[z]+b\,\text{Sinh}[z])}$

- **Rule: If** $a^2-b^2\neq 0 \land m,\,n,\,p\in\mathbb{Z} \land m>0 \land n>0 \land p<0$**, then**

$$\int \text{Cosh}[c+d\,x]^m\,\text{Sinh}[c+d\,x]^n\,(a\,\text{Cosh}[c+d\,x]+b\,\text{Sinh}[c+d\,x])^p\,dx \longrightarrow$$
$$-\frac{b}{a^2-b^2}\int \text{Cosh}[c+d\,x]^m\,\text{Sinh}[c+d\,x]^{n-1}\,(a\,\text{Cosh}[c+d\,x]+b\,\text{Sinh}[c+d\,x])^{p+1}\,dx\,+$$
$$\frac{a}{a^2-b^2}\int \text{Cosh}[c+d\,x]^{m-1}\,\text{Sinh}[c+d\,x]^n\,(a\,\text{Cosh}[c+d\,x]+b\,\text{Sinh}[c+d\,x])^{p+1}\,dx\,+$$
$$\frac{a\,b}{a^2-b^2}\int \text{Cosh}[c+d\,x]^{m-1}\,\text{Sinh}[c+d\,x]^{n-1}\,(a\,\text{Cosh}[c+d\,x]+b\,\text{Sinh}[c+d\,x])^p\,dx$$

- **Program code:**

```
Int[Cosh[c_.+d_.*x_]^m_.*Sinh[c_.+d_.*x_]^n_.*(a_.*Cosh[c_.+d_.*x_]+b_.*Sinh[c_.+d_.*x_])^p_,x_Symbo
  -Dist[b/(a^2-b^2),Int[Cosh[c+d*x]^m*Sinh[c+d*x]^(n-1)*(a*Cosh[c+d*x]+b*Sinh[c+d*x])^(p+1),x]] +
  Dist[a/(a^2-b^2),Int[Cosh[c+d*x]^(m-1)*Sinh[c+d*x]^n*(a*Cosh[c+d*x]+b*Sinh[c+d*x])^(p+1),x]] +
  Dist[a*b/(a^2-b^2),Int[Cosh[c+d*x]^(m-1)*Sinh[c+d*x]^(n-1)*(a*Cosh[c+d*x]+b*Sinh[c+d*x])^p,x]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2-b^2] && IntegersQ[m,n,p] && m>0 && n>0 && p<0
```

- **Derivation: Algebraic expansion**

- **Basis:** $\frac{\text{Sinh}[z]^2}{a\,\text{Cosh}[z]+b\,\text{Sinh}[z]} = -\frac{b\,\text{Sinh}[z]}{a^2-b^2} + \frac{a\,\text{Cosh}[z]}{a^2-b^2} - \frac{a^2}{(a^2-b^2)\,(a\,\text{Cosh}[z]+b\,\text{Sinh}[z])}$

- **Rule: If** $a^2-b^2\neq 0 \land n\in\mathbb{Z} \land n>1$**, then**

$$\int \frac{u\,\text{Sinh}[c+d\,x]^n}{a\,\text{Cosh}[c+d\,x]+b\,\text{Sinh}[c+d\,x]}\,dx \longrightarrow -\frac{b}{a^2-b^2}\int u\,\text{Sinh}[c+d\,x]^{n-1}\,dx\,+$$
$$\frac{a}{a^2-b^2}\int u\,\text{Sinh}[c+d\,x]^{n-2}\,\text{Cosh}[c+d\,x]\,dx - \frac{a^2}{a^2-b^2}\int \frac{u\,\text{Sinh}[c+d\,x]^{n-2}}{a\,\text{Cosh}[c+d\,x]+b\,\text{Sinh}[c+d\,x]}\,dx$$

- **Program code:**

```
Int[u_.*Sinh[c_.+d_.*x_]^n_/(a_.*Cosh[c_.+d_.*x_]+b_.*Sinh[c_.+d_.*x_]),x_Symbol] :=
  Dist[-b/(a^2-b^2),Int[u*Sinh[c+d*x]^(n-1),x]] +
  Dist[a/(a^2-b^2),Int[u*Sinh[c+d*x]^(n-2)*Cosh[c+d*x],x]] -
  Dist[a^2/(a^2-b^2),Int[u*Sinh[c+d*x]^(n-2)/(a*Cosh[c+d*x]+b*Sinh[c+d*x]),x]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2-b^2] && IntegerQ[n] && n>0 &&
(n>1 || MatchQ[u,v_.*Tanh[c+d*x]^m_. /; IntegerQ[m] && m>0])
```

- **Derivation: Algebraic expansion**

- **Basis:** $\frac{\text{Cosh}[z]^2}{a\,\text{Cosh}[z]+b\,\text{Sinh}[z]} = \frac{a\,\text{Cosh}[z]}{a^2-b^2} - \frac{b\,\text{Sinh}[z]}{a^2-b^2} - \frac{b^2}{(a^2-b^2)\,(a\,\text{Cosh}[z]+b\,\text{Sinh}[z])}$

- **Rule: If** $a^2 - b^2 \neq 0 \land n \in \mathbb{Z} \land n > 1$**, then**

$$\int \frac{u\,\text{Cosh}[c+d\,x]^n}{a\,\text{Cosh}[c+d\,x]+b\,\text{Sinh}[c+d\,x]}\,dx \longrightarrow \frac{a}{a^2-b^2}\int u\,\text{Cosh}[c+d\,x]^{n-1}\,dx -$$
$$\frac{b}{a^2-b^2}\int u\,\text{Cosh}[c+d\,x]^{n-2}\,\text{Sinh}[c+d\,x]\,dx - \frac{b^2}{a^2-b^2}\int \frac{u\,\text{Cosh}[c+d\,x]^{n-2}}{a\,\text{Cosh}[c+d\,x]+b\,\text{Sinh}[c+d\,x]}\,dx$$

- **Program code:**

```
Int[u_.*Cosh[c_.+d_.*x_]^n_/(a_.*Cosh[c_.+d_.*x_]+b_.*Sinh[c_.+d_.*x_]),x_Symbol] :=
  Dist[a/(a^2-b^2),Int[u*Cosh[c+d*x]^(n-1),x]] -
  Dist[b/(a^2-b^2),Int[u*Cosh[c+d*x]^(n-2)*Sinh[c+d*x],x]] -
  Dist[b^2/(a^2-b^2),Int[u*Cosh[c+d*x]^(n-2)/(a*Cosh[c+d*x]+b*Sinh[c+d*x]),x]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a^2-b^2] && IntegerQ[n] && n>0 &&
(n>1 || MatchQ[u,v_.*Coth[c+d*x]^m_. /; IntegerQ[m] && m>0])
```

$$\int \frac{1}{a + b\,\text{Cosh}[d + e\,x] + c\,\text{Sinh}[d + e\,x]}\,dx$$

- **Reference:** G&R 2.451.4c
- **Rule:** If $a - b = 0$, then

$$\int \frac{1}{a + b\,\text{Cosh}[d + e\,x] + c\,\text{Sinh}[d + e\,x]}\,dx \longrightarrow \frac{1}{c\,e}\,\text{Log}\left[a + c\,\text{Tanh}\left[\frac{d + e\,x}{2}\right]\right]$$

- **Program code:**

```
Int[1/(a_+b_.*Cosh[d_.+e_.*x_]+c_.*Sinh[d_.+e_.*x_]),x_Symbol] :=
  Log[a+c*Tanh[(d+e*x)/2]]/(c*e) /;
FreeQ[{a,b,c,d,e},x] && ZeroQ[a-b]
```

- **Reference:** G&R 2.451.4c
- **Rule:** If $a + b = 0$, then

$$\int \frac{1}{a + b\,\text{Cosh}[d + e\,x] + c\,\text{Sinh}[d + e\,x]}\,dx \longrightarrow -\frac{1}{c\,e}\,\text{Log}\left[a - c\,\text{Coth}\left[\frac{d + e\,x}{2}\right]\right]$$

- **Program code:**

```
Int[1/(a_+b_.*Cosh[d_.+e_.*x_]+c_.*Sinh[d_.+e_.*x_]),x_Symbol] :=
  -Log[a-c*Coth[(d+e*x)/2]]/(c*e) /;
FreeQ[{a,b,c,d,e},x] && ZeroQ[a+b]
```

- **Reference:** G&R 2.451.4d
- **Rule:** If $a^2 - b^2 + c^2 = 0$, then

$$\int \frac{1}{a + b\,\text{Cosh}[d + e\,x] + c\,\text{Sinh}[d + e\,x]}\,dx \longrightarrow -\frac{c + a\,\text{Sinh}[d + e\,x]}{c\,e\,(c\,\text{Cosh}[d + e\,x] + b\,\text{Sinh}[d + e\,x])}$$

- **Program code:**

```
Int[1/(a_+b_.*Cosh[d_.+e_.*x_]+c_.*Sinh[d_.+e_.*x_]),x_Symbol] :=
  -(c+a*Sinh[d+e*x])/(c*e*(c*Cosh[d+e*x]+b*Sinh[d+e*x])) /;
FreeQ[{a,b,c,d,e},x] && ZeroQ[a^2-b^2+c^2]
```

- **Reference: G&R 2.451.4b'**
- **Rule: If** $a^2 - b^2 \neq 0 \wedge a^2 - b^2 + c^2 > 0$, **then**

$$\int \frac{1}{a + b\,\mathrm{Cosh}[d + e\,x] + c\,\mathrm{Sinh}[d + e\,x]}\,dx \longrightarrow -\frac{2}{e\sqrt{a^2 - b^2 + c^2}}\,\mathrm{ArcTanh}\Big[\frac{c - (a - b)\,\mathrm{Tanh}\big[\frac{d+e\,x}{2}\big]}{\sqrt{a^2 - b^2 + c^2}}\Big]$$

- **Program code:**

```
Int[1/(a_.+b_.*Cosh[d_.+e_.*x_]+c_.*Sinh[d_.+e_.*x_]),x_Symbol] :=
  -2*ArcTanh[(c-(a-b)*Tanh[(d+e*x)/2])/Rt[a^2-b^2+c^2,2]]/(e*Rt[a^2-b^2+c^2,2]) /;
FreeQ[{a,b,c,d,e},x] && NonzeroQ[a^2-b^2] && PosQ[a^2-b^2+c^2]
```

- **Reference: Reference: G&R 2.451.4a**
- **Rule: If** $a^2 - b^2 \neq 0 \wedge \neg\,(a^2 - b^2 + c^2 > 0)$, **then**

$$\int \frac{1}{a + b\,\mathrm{Cosh}[d + e\,x] + c\,\mathrm{Sinh}[d + e\,x]}\,dx \longrightarrow \frac{2}{e\sqrt{-a^2 + b^2 - c^2}}\,\mathrm{ArcTan}\Big[\frac{c - (a - b)\,\mathrm{Tanh}\big[\frac{d+e\,x}{2}\big]}{\sqrt{-a^2 + b^2 - c^2}}\Big]$$

- **Program code:**

```
Int[1/(a_.+b_.*Cosh[d_.+e_.*x_]+c_.*Sinh[d_.+e_.*x_]),x_Symbol] :=
  2*ArcTan[(c-(a-b)*Tanh[(d+e*x)/2])/Rt[-a^2+b^2-c^2,2]]/(e*Rt[-a^2+b^2-c^2,2]) /;
FreeQ[{a,b,c,d,e},x] && NonzeroQ[a^2-b^2] && NegQ[a^2-b^2+c^2]
```

$$\int \sqrt{a + b\,\text{Cosh}[d + e\,x] + c\,\text{Sinh}[d + e\,x]}\,dx$$

- **Rule: If** $a^2 - b^2 + c^2 = 0$, **then**

$$\int \sqrt{a + b\,\text{Cosh}[d + e\,x] + c\,\text{Sinh}[d + e\,x]}\,dx \longrightarrow \frac{2\,(c\,\text{Cosh}[d + e\,x] + b\,\text{Sinh}[d + e\,x])}{e\,\sqrt{a + b\,\text{Cosh}[d + e\,x] + c\,\text{Sinh}[d + e\,x]}}$$

- **Program code:**

```
Int[Sqrt[a_+b_.*Cosh[d_.+e_.*x_]+c_.*Sinh[d_.+e_.*x_]],x_Symbol] :=
  2*(c*Cosh[d+e*x]+b*Sinh[d+e*x])/(e*Sqrt[a+b*Cosh[d+e*x]+c*Sinh[d+e*x]]) /;
FreeQ[{a,b,c,d,e},x] && ZeroQ[a^2-b^2+c^2]
```

- **Derivation: Algebraic simplification**

- **Basis:** $a + b\,\text{Cosh}[z] + c\,\text{Sinh}[z] = a - i\,\sqrt{b^2 - c^2}\,\text{Sinh}[z + i\,\text{ArcTan}[i\,c,\,b]]$

- **Rule: If** $a^2 - b^2 + c^2 \neq 0 \bigwedge a - \sqrt{b^2 - c^2} > 0$, **then**

$$\int \sqrt{a + b\,\text{Cosh}[d + e\,x] + c\,\text{Sinh}[d + e\,x]}\,dx \longrightarrow \int \sqrt{a - i\,\sqrt{b^2 - c^2}\,\text{Sinh}[d + e\,x + i\,\text{ArcTan}[i\,c,\,b]]}\,dx$$

- **Program code:**

```
Int[Sqrt[a_.+b_.*Cosh[d_.+e_.*x_]+c_.*Sinh[d_.+e_.*x_]],x_Symbol] :=
  Int[Sqrt[a-I*Sqrt[b^2-c^2]*Sinh[d+e*x+I*ArcTan[I*c,b]]],x] /;
FreeQ[{a,b,c,d,e},x] && NonzeroQ[a^2-b^2+c^2] && PositiveQ[a-Sqrt[b^2-c^2]]
```

- **Derivation: Algebraic simplification**

- **Basis:** $a+b\,\mathrm{Cosh}[z]+c\,\mathrm{Sinh}[z]=a-i\sqrt{b^2-c^2}\,\mathrm{Sinh}[z+i\,\mathrm{ArcTan}[i\,c,\,b]]$

- **Rule: If** $a^2-b^2+c^2\neq 0\ \bigwedge\ \neg\left(a-\sqrt{b^2-c^2}>0\right)$, **then**

$$\int\sqrt{a+b\,\mathrm{Cosh}[d+e\,x]+c\,\mathrm{Sinh}[d+e\,x]}\,dx\ \longrightarrow$$

$$\frac{2\,i\sqrt{a+b\,\mathrm{Cosh}[d+e\,x]+c\,\mathrm{Sinh}[d+e\,x]}}{e\sqrt{\frac{a+b\,\mathrm{Cosh}[d+e\,x]+c\,\mathrm{Sinh}[d+e\,x]}{a-\sqrt{b^2-c^2}}}}\,\mathrm{EllipticE}\left[\frac{1}{2}\left(\frac{\pi}{2}-i\,(d+e\,x+i\,\mathrm{ArcTan}[i\,c,\,b])\right),\ \frac{2}{1-\frac{a}{\sqrt{b^2-c^2}}}\right]$$

- **Program code:**

```
Int[Sqrt[a_.+b_.*Cosh[d_.+e_.*x_]+c_.*Sinh[d_.+e_.*x_]],x_Symbol] :=
  2*I*EllipticE[(Pi/2-I*(d+e*x+I*ArcTan[I*c,b]))/2,2/(1-a/Sqrt[b^2-c^2])]*
  Sqrt[a+b*Cosh[d+e*x]+c*Sinh[d+e*x]]/
  (e*Sqrt[(a+b*Cosh[d+e*x]+c*Sinh[d+e*x])/(a-Sqrt[b^2-c^2])]) /;
FreeQ[{a,b,c,d,e},x] && NonzeroQ[a^2-b^2+c^2] && Not[PositiveQ[a-Sqrt[b^2-c^2]]]
```

$$\int \frac{1}{\sqrt{a + b\,\mathrm{Cosh}[d + e\,x] + c\,\mathrm{Sinh}[d + e\,x]}}\,dx$$

- **Derivation: Algebraic simplification**
- **Basis:** $a + b\,\mathrm{Cosh}[z] + c\,\mathrm{Sinh}[z] = a - i\sqrt{b^2 - c^2}\,\mathrm{Sinh}[z + i\,\mathrm{ArcTan}[i\,c, b]]$
- **Rule: If** $a - \sqrt{b^2 - c^2} > 0$**, then**

$$\int \frac{1}{\sqrt{a + b\,\mathrm{Cosh}[d + e\,x] + c\,\mathrm{Sinh}[d + e\,x]}}\,dx \longrightarrow \int \frac{1}{\sqrt{a - i\sqrt{b^2 - c^2}\,\mathrm{Sinh}[d + e\,x + i\,\mathrm{ArcTan}[i\,c, b]]}}\,dx$$

- **Program code:**

```
Int[1/Sqrt[a_.+b_.*Cosh[d_.+e_.*x_]+c_.*Sinh[d_.+e_.*x_]],x_Symbol] :=
  Int[1/Sqrt[a-I*Sqrt[b^2-c^2]*Sinh[d+e*x+I*ArcTan[I*c,b]]],x] /;
FreeQ[{a,b,c,d,e},x] && PositiveQ[a-Sqrt[b^2-c^2]]
```

- **Derivation: Piecewise constant extraction and algebraic simplification**
- **Basis:** $\partial_z \frac{\sqrt{\frac{a+b\,\mathrm{Cosh}[z]+c\,\mathrm{Sinh}[z]}{a-\sqrt{b^2-c^2}}}}{\sqrt{a+b\,\mathrm{Cosh}[z]+c\,\mathrm{Sinh}[z]}} = 0$
- **Basis:** $a + b\,\mathrm{Cosh}[z] + c\,\mathrm{Sinh}[z] = a - i\sqrt{b^2 - c^2}\,\mathrm{Sinh}[z + i\,\mathrm{ArcTan}[i\,c, b]]$
- **Rule: If** $a - \sqrt{b^2 - c^2} \neq 0 \bigwedge \neg\left(a - \sqrt{b^2 - c^2} > 0\right)$**, then**

$$\int \frac{1}{\sqrt{a + b\,\mathrm{Cosh}[d + e\,x] + c\,\mathrm{Sinh}[d + e\,x]}}\,dx \longrightarrow$$

$$\frac{2\,i\sqrt{\frac{a+b\,\mathrm{Cosh}[d+e\,x]+c\,\mathrm{Sinh}[d+e\,x]}{a-\sqrt{b^2-c^2}}}}{e\sqrt{a + b\,\mathrm{Cosh}[d + e\,x] + c\,\mathrm{Sinh}[d + e\,x]}}\,\mathrm{EllipticF}\left[\frac{1}{2}\left(\frac{\pi}{2} - i\,(d + e\,x + i\,\mathrm{ArcTan}[i\,c, b])\right), \frac{2}{1 - \frac{a}{\sqrt{b^2-c^2}}}\right]$$

- **Program code:**

```
Int[1/Sqrt[a_.+b_.*Cosh[d_.+e_.*x_]+c_.*Sinh[d_.+e_.*x_]],x_Symbol] :=
  2*I*EllipticF[(Pi/2-I*(d+e*x+I*ArcTan[I*c,b]))/2,2/(1-a/Sqrt[b^2-c^2])]*
  Sqrt[(a+b*Cosh[d+e*x]+c*Sinh[d+e*x])/(a-Sqrt[b^2-c^2])]/
  (e*Sqrt[a+b*Cosh[d+e*x]+c*Sinh[d+e*x]]) /;
FreeQ[{a,b,c,d,e},x] && NonzeroQ[a-Sqrt[b^2-c^2]] && Not[PositiveQ[a-Sqrt[b^2-c^2]]]
```

$$\int (a + b \operatorname{Cosh}[d + e x] + c \operatorname{Sinh}[d + e x])^n \, dx$$

- **Reference: G&R 2.451.1 inverted with $a^2 - b^2 + c^2 = 0$**
- **Rule: If $a^2 - b^2 + c^2 = 0 \wedge n > 1$, then**

$$\int (a + b \operatorname{Cosh}[d + e x] + c \operatorname{Sinh}[d + e x])^n \, dx \longrightarrow$$
$$\frac{(c \operatorname{Cosh}[d + e x] + b \operatorname{Sinh}[d + e x]) (a + b \operatorname{Cosh}[d + e x] + c \operatorname{Sinh}[d + e x])^{n-1}}{e n} +$$
$$\frac{a (2 n - 1)}{n} \int (a + b \operatorname{Cosh}[d + e x] + c \operatorname{Sinh}[d + e x])^{n-1} \, dx$$

- **Program code:**

```
Int[(a_+b_.*Cosh[d_.+e_.*x_]+c_.*Sinh[d_.+e_.*x_])^n_,x_Symbol] :=
  (c*Cosh[d+e*x]+b*Sinh[d+e*x])*(a+b*Cosh[d+e*x]+c*Sinh[d+e*x])^(n-1)/(e*n) +
  Dist[a*(2*n-1)/n,Int[(a+b*Cosh[d+e*x]+c*Sinh[d+e*x])^(n-1),x]] /;
FreeQ[{a,b,c,d,e},x] && ZeroQ[a^2-b^2+c^2] && RationalQ[n] && n>1
```

- **Reference: G&R 2.451.1 inverted**
- **Rule: If $a^2 - b^2 + c^2 \neq 0 \wedge n > 1$, then**

$$\int (a + b \operatorname{Cosh}[d + e x] + c \operatorname{Sinh}[d + e x])^n \, dx \longrightarrow$$
$$\frac{(c \operatorname{Cosh}[d + e x] + b \operatorname{Sinh}[d + e x]) (a + b \operatorname{Cosh}[d + e x] + c \operatorname{Sinh}[d + e x])^{n-1}}{e n} +$$
$$\frac{1}{n} \int \left(n a^2 + (n - 1) \left(b^2 - c^2\right) + a b (2 n - 1) \operatorname{Cosh}[d + e x] + a c (2 n - 1) \operatorname{Sinh}[d + e x]\right)$$
$$(a + b \operatorname{Cosh}[d + e x] + c \operatorname{Sinh}[d + e x])^{n-2} \, dx$$

- **Program code:**

```
Int[(a_+b_.*Cosh[d_.+e_.*x_]+c_.*Sinh[d_.+e_.*x_])^n_,x_Symbol] :=
  (c*Cosh[d+e*x]+b*Sinh[d+e*x])*(a+b*Cosh[d+e*x]+c*Sinh[d+e*x])^(n-1)/(e*n) +
  Dist[1/n,Int[(n*a^2+(n-1)*(b^2-c^2)+a*b*(2*n-1)*Cosh[d+e*x]+a*c*(2*n-1)*Sinh[d+e*x])*
    (a+b*Cosh[d+e*x]+c*Sinh[d+e*x])^(n-2),x]] /;
FreeQ[{a,b,c,d,e},x] && NonzeroQ[a^2-b^2+c^2] && RationalQ[n] && n>1
```

$$\int \frac{1}{(a + b\,\mathrm{Cosh}[d + e\,x] + c\,\mathrm{Sinh}[d + e\,x])^n}\,dx$$

- **Rule: If** $a^2 - b^2 + c^2 = 0 \wedge n < -1$, **then**

$$\int (a + b\,\mathrm{Cosh}[d + e\,x] + c\,\mathrm{Sinh}[d + e\,x])^n\,dx \longrightarrow$$
$$-\frac{(c\,\mathrm{Cosh}[d + e\,x] + b\,\mathrm{Sinh}[d + e\,x])\,(a + b\,\mathrm{Cosh}[d + e\,x] + c\,\mathrm{Sinh}[d + e\,x])^n}{a\,e\,(2\,n + 1)} +$$
$$\frac{n + 1}{a\,(2\,n + 1)} \int (a + b\,\mathrm{Cosh}[d + e\,x] + c\,\mathrm{Sinh}[d + e\,x])^{n+1}\,dx$$

- **Program code:**

```
Int[(a_+b_.*Cosh[d_.+e_.*x_]+c_.*Sinh[d_.+e_.*x_])^n_,x_Symbol] :=
  -(c*Cosh[d+e*x]+b*Sinh[d+e*x])*(a+b*Cosh[d+e*x]+c*Sinh[d+e*x])^n/(a*e*(2*n+1)) +
  Dist[(n+1)/(a*(2*n+1)),Int[(a+b*Cosh[d+e*x]+c*Sinh[d+e*x])^(n+1),x]] /;
FreeQ[{a,b,c,d,e},x] && ZeroQ[a^2-b^2+c^2] && RationalQ[n] && n<-1
```

- **Reference: G&R 2.451.1 with** $n = -2$
- **Rule: If** $a^2 - b^2 + c^2 \neq 0$, **then**

$$\int \frac{1}{(a + b\,\mathrm{Cosh}[d + e\,x] + c\,\mathrm{Sinh}[d + e\,x])^2}\,dx \longrightarrow$$
$$-\frac{c\,\mathrm{Cosh}[d + e\,x] + b\,\mathrm{Sinh}[d + e\,x]}{e\,\left(a^2 - b^2 + c^2\right)\,(a + b\,\mathrm{Cosh}[d + e\,x] + c\,\mathrm{Sinh}[d + e\,x])} +$$
$$\frac{a}{a^2 - b^2 + c^2} \int \frac{1}{a + b\,\mathrm{Cosh}[d + e\,x] + c\,\mathrm{Sinh}[d + e\,x]}\,dx$$

- **Program code:**

```
Int[1/(a_+b_.*Cosh[d_.+e_.*x_]+c_.*Sinh[d_.+e_.*x_])^2,x_Symbol] :=
  -(c*Cosh[d+e*x]+b*Sinh[d+e*x])/(e*(a^2-b^2+c^2)*(a+b*Cosh[d+e*x]+c*Sinh[d+e*x])) +
  Dist[a/(a^2-b^2+c^2),Int[1/(a+b*Cosh[d+e*x]+c*Sinh[d+e*x]),x]] /;
FreeQ[{a,b,c,d,e},x] && NonzeroQ[a^2-b^2+c^2]
```

- **Reference: G&R 2.451.1 with** $n = -\frac{3}{2}$
- **Rule: If** $a^2 - b^2 + c^2 \neq 0$**, then**

$$\int \frac{1}{(a + b \operatorname{Cosh}[d + e x] + c \operatorname{Sinh}[d + e x])^{3/2}} \, dx \longrightarrow$$

$$-\frac{2 (c \operatorname{Cosh}[d + e x] + b \operatorname{Sinh}[d + e x])}{e \left(a^2 - b^2 + c^2\right) \sqrt{a + b \operatorname{Cosh}[d + e x] + c \operatorname{Sinh}[d + e x]}} + \frac{1}{a^2 - b^2 + c^2} \int \sqrt{a + b \operatorname{Cosh}[d + e x] + c \operatorname{Sinh}[d + e x]} \, dx$$

- **Program code:**

```
Int[1/(a_+b_.*Cosh[d_.+e_.*x_]+c_.*Sinh[d_.+e_.*x_])^(3/2),x_Symbol] :=
  -2*(c*Cosh[d+e*x]+b*Sinh[d+e*x])/(e*(a^2-b^2+c^2)*Sqrt[a+b*Cosh[d+e*x]+c*Sinh[d+e*x]]) +
  Dist[1/(a^2-b^2+c^2),Int[Sqrt[a+b*Cosh[d+e*x]+c*Sinh[d+e*x]],x]] /;
FreeQ[{a,b,c,d,e},x] && NonzeroQ[a^2-b^2+c^2]
```

- **Reference: G&R 2.451.1**
- **Rule: If** $a^2 - b^2 + c^2 \neq 0 \bigwedge n < -1 \bigwedge n \neq -2 \bigwedge n \neq -\frac{3}{2}$**, then**

$$\int (a + b \operatorname{Cosh}[d + e x] + c \operatorname{Sinh}[d + e x])^n \, dx \longrightarrow$$

$$\frac{(c \operatorname{Cosh}[d + e x] + b \operatorname{Sinh}[d + e x]) (a + b \operatorname{Cosh}[d + e x] + c \operatorname{Sinh}[d + e x])^{n+1}}{e (n + 1) \left(a^2 - b^2 + c^2\right)} + \frac{1}{(n + 1) \left(a^2 - b^2 + c^2\right)}$$

$$\int ((n + 1) a - (n + 2) b \operatorname{Cosh}[d + e x] - (n + 2) c \operatorname{Sinh}[d + e x]) (a + b \operatorname{Cosh}[d + e x] + c \operatorname{Sinh}[d + e x])^{n+1} \, dx$$

- **Program code:**

```
Int[(a_+b_.*Cosh[d_.+e_.*x_]+c_.*Sinh[d_.+e_.*x_])^n_,x_Symbol] :=
  (c*Cosh[d+e*x]+b*Sinh[d+e*x])*(a+b*Cosh[d+e*x]+c*Sinh[d+e*x])^(n+1)/(e*(n+1)*(a^2-b^2+c^2)) +
  1/((n+1)*(a^2-b^2+c^2))*
    Int[((n+1)*a-(n+2)*b*Cosh[d+e*x]-(n+2)*c*Sinh[d+e*x])*(a+b*Cosh[d+e*x]+c*Sinh[d+e*x])^(n+1),x] /
FreeQ[{a,b,c,d,e},x] && NonzeroQ[a^2-b^2+c^2] && RationalQ[n] && n<-1 && n≠-2 && n≠-3/2
```

$$\int (A + B\,\text{Cosh}[d + e\,x] + C\,\text{Sinh}[d + e\,x])\,(a + b\,\text{Cosh}[d + e\,x] + c\,\text{Sinh}[d + e\,x])^n\,dx$$

- **Reference: G&R 2.451.3**
- **Rule: If** $b^2 - c^2 = 0$**, then**

$$\int \frac{A + B\,\text{Cosh}[d + e\,x] + C\,\text{Sinh}[d + e\,x]}{a + b\,\text{Cosh}[d + e\,x] + c\,\text{Sinh}[d + e\,x]}\,dx \longrightarrow \frac{(2\,a\,A - b\,B + c\,C)\,x}{2\,a^2} - \frac{(b\,B - c\,C)\,(b\,\text{Cosh}[d + e\,x] - c\,\text{Sinh}[d + e\,x])}{2\,a\,b\,c\,e} + \frac{\left(a^2\,(b\,B + c\,C) - 2\,a\,A\,b^2 + b^2\,(b\,B - c\,C)\right)\,\text{Log}[a + b\,\text{Cosh}[d + e\,x] + c\,\text{Sinh}[d + e\,x]]}{2\,a^2\,b\,c\,e}$$

- **Program code:**

```
Int[(A_.+B_.*Cosh[d_.+e_.*x_]+C_.*Sinh[d_.+e_.*x_])/(a_+b_.*Cosh[d_.+e_.*x_]+c_.*Sinh[d_.+e_.*x_]),x
  (2*a*A-b*B+c*C)*x/(2*a^2) - (b*B-c*C)*(b*Cosh[d+e*x]-c*Sinh[d+e*x])/(2*a*b*c*e) +
  (a^2*(b*B+c*C)-2*a*A*b^2+b^2*(b*B-c*C))*Log[a+b*Cosh[d+e*x]+c*Sinh[d+e*x]]/(2*a^2*b*c*e) /;
FreeQ[{a,b,c,d,e,A,B,C},x] && ZeroQ[b^2-c^2]
```

- **Reference: G&R 2.451.3 with** $B = 0$

```
Int[(A_.+C_.*Sinh[d_.+e_.*x_])/(a_+b_.*Cosh[d_.+e_.*x_]+c_.*Sinh[d_.+e_.*x_]),x_Symbol] :=
  (2*a*A+c*C)*x/(2*a^2) + C*Cosh[d+e*x]/(2*a*e) - c*C*Sinh[d+e*x]/(2*a*b*e) +
  (a^2*C-2*a*c*A-b^2*C)*Log[a+b*Cosh[d+e*x]+c*Sinh[d+e*x]]/(2*a^2*b*e) /;
FreeQ[{a,b,c,d,e,A,C},x] && ZeroQ[b^2-c^2]
```

- **Reference: G&R 2.451.3 with** $C = 0$

```
Int[(A_.+B_.*Cosh[d_.+e_.*x_])/(a_+b_.*Cosh[d_.+e_.*x_]+c_.*Sinh[d_.+e_.*x_]),x_Symbol] :=
  (2*a*A-b*B)*x/(2*a^2) - b*B*Cosh[d+e*x]/(2*a*c*e) + B*Sinh[d+e*x]/(2*a*e) +
  (a^2*B-2*a*b*A+b^2*B)*Log[a+b*Cosh[d+e*x]+c*Sinh[d+e*x]]/(2*a^2*c*e) /;
FreeQ[{a,b,c,d,e,A,B},x] && ZeroQ[b^2-c^2]
```

- **Reference: G&R 2.451.2 with** $A\,(b^2-c^2)-a\,(b\,B-c\,C)=0$

- **Rule: If** $b^2-c^2\neq 0\,\bigwedge\,A\,(b^2-c^2)-a\,(b\,B-c\,C)=0$**, then**

$$\int\frac{A+B\,\mathrm{Cosh}[d+e\,x]+C\,\mathrm{Sinh}[d+e\,x]}{a+b\,\mathrm{Cosh}[d+e\,x]+c\,\mathrm{Sinh}[d+e\,x]}\,dx\;\longrightarrow\;\frac{(b\,B-c\,C)\,x}{b^2-c^2}-\frac{(c\,B-b\,C)\,\mathrm{Log}[a+b\,\mathrm{Cosh}[d+e\,x]+c\,\mathrm{Sinh}[d+e\,x]]}{e\,(b^2-c^2)}$$

- **Program code:**

```
Int[(A_.+B_.*Cosh[d_.+e_.*x_]+C_.*Sinh[d_.+e_.*x_])/(a_.+b_.*Cosh[d_.+e_.*x_]+c_.*Sinh[d_.+e_.*x_]),
  (b*B-c*C)*x/(b^2-c^2) - (c*B-b*C)*Log[a+b*Cosh[d+e*x]+c*Sinh[d+e*x]]/(e*(b^2-c^2)) /;
FreeQ[{a,b,c,d,e,A,B,C},x] && NonzeroQ[b^2-c^2] && ZeroQ[A*(b^2-c^2)-a*(b*B-c*C)]
```

- **Reference: G&R 2.451.2 with** $B=0$ **and** $A\,(b^2-c^2)+a\,c\,C=0$

```
Int[(A_.+C_.*Sinh[d_.+e_.*x_])/(a_.+b_.*Cosh[d_.+e_.*x_]+c_.*Sinh[d_.+e_.*x_]),x_Symbol] :=
  -c*C*x/(b^2-c^2) + b*C*Log[a+b*Cosh[d+e*x]+c*Sinh[d+e*x]]/(e*(b^2-c^2)) /;
FreeQ[{a,b,c,d,e,A,C},x] && NonzeroQ[b^2-c^2] && ZeroQ[A*(b^2-c^2)+a*c*C]
```

- **Reference: G&R 2.451.2 with** $C=0$ **and** $A\,(b^2-c^2)-a\,b\,B=0$

```
Int[(A_.+B_.*Cosh[d_.+e_.*x_])/(a_.+b_.*Cosh[d_.+e_.*x_]+c_.*Sinh[d_.+e_.*x_]),x_Symbol] :=
  b*B*x/(b^2-c^2) - c*B*Log[a+b*Cosh[d+e*x]+c*Sinh[d+e*x]]/(e*(b^2-c^2)) /;
FreeQ[{a,b,c,d,e,A,B},x] && NonzeroQ[b^2-c^2] && ZeroQ[A*(b^2-c^2)-a*b*B]
```

- **Reference: G&R 2.451.2**

- **Rule: If** $b^2-c^2\neq 0\,\bigwedge\,A\,(b^2-c^2)-a\,(b\,B-c\,C)\neq 0$**, then**

$$\int\frac{A+B\,\mathrm{Cosh}[d+e\,x]+C\,\mathrm{Sinh}[d+e\,x]}{a+b\,\mathrm{Cosh}[d+e\,x]+c\,\mathrm{Sinh}[d+e\,x]}\,dx\;\longrightarrow$$
$$\frac{(b\,B-c\,C)\,x}{b^2-c^2}-\frac{(c\,B-b\,C)\,\mathrm{Log}[a+b\,\mathrm{Cosh}[d+e\,x]+c\,\mathrm{Sinh}[d+e\,x]]}{e\,(b^2-c^2)}+$$
$$\frac{A\,(b^2-c^2)-a\,(b\,B-c\,C)}{b^2-c^2}\int\frac{1}{a+b\,\mathrm{Cosh}[d+e\,x]+c\,\mathrm{Sinh}[d+e\,x]}\,dx$$

- **Program code:**

```
Int[(A_.+B_.*Cosh[d_.+e_.*x_]+C_.*Sinh[d_.+e_.*x_])/(a_.+b_.*Cosh[d_.+e_.*x_]+c_.*Sinh[d_.+e_.*x_]),
  (b*B-c*C)*x/(b^2-c^2) - (c*B-b*C)*Log[a+b*Cosh[d+e*x]+c*Sinh[d+e*x]]/(e*(b^2-c^2)) +
  Dist[(A*(b^2-c^2)-a*(b*B-c*C))/(b^2-c^2),Int[1/(a+b*Cosh[d+e*x]+c*Sinh[d+e*x]),x]] /;
FreeQ[{a,b,c,d,e,A,B,C},x] && NonzeroQ[b^2-c^2] && NonzeroQ[A*(b^2-c^2)-a*(b*B-c*C)]
```

- **Reference: G&R 2.451.2 with $B = 0$**

```
Int[(A_.+C_.*Sinh[d_.+e_.*x_])/(a_.+b_.*Cosh[d_.+e_.*x_]+c_.*Sinh[d_.+e_.*x_]),x_Symbol] :=
  -c*C*x/(b^2-c^2) + b*C*Log[a+b*Cosh[d+e*x]+c*Sinh[d+e*x]]/(e*(b^2-c^2)) +
  Dist[(A*(b^2-c^2)+a*c*C)/(b^2-c^2),Int[1/(a+b*Cosh[d+e*x]+c*Sinh[d+e*x]),x]] /;
FreeQ[{a,b,c,d,e,A,C},x] && NonzeroQ[b^2-c^2] && NonzeroQ[A*(b^2-c^2)+a*c*C]
```

- **Reference: G&R 2.451.2 with $C = 0$**

```
Int[(A_.+B_.*Cosh[d_.+e_.*x_])/(a_.+b_.*Cosh[d_.+e_.*x_]+c_.*Sinh[d_.+e_.*x_]),x_Symbol] :=
  b*B*x/(b^2-c^2) - c*B*Log[a+b*Cosh[d+e*x]+c*Sinh[d+e*x]]/(e*(b^2-c^2)) +
  Dist[(A*(b^2-c^2)-a*b*B)/(b^2-c^2),Int[1/(a+b*Cosh[d+e*x]+c*Sinh[d+e*x]),x]] /;
FreeQ[{a,b,c,d,e,A,B},x] && NonzeroQ[b^2-c^2] && NonzeroQ[A*(b^2-c^2)-a*b*B]
```

- **Reference: G&R 2.451.1 with $n = -2$ and $a A - b B + c C = 0$**

- **Rule: If $a^2 - b^2 + c^2 \neq 0 \wedge a A - b B + c C = 0$, then**

$$\int \frac{A + B \operatorname{Cosh}[d + e x] + C \operatorname{Sinh}[d + e x]}{(a + b \operatorname{Cosh}[d + e x] + c \operatorname{Sinh}[d + e x])^2} \, dx \longrightarrow -\frac{c B - b C - (a C - c A) \operatorname{Cosh}[d + e x] + (b A - a B) \operatorname{Sinh}[d + e x]}{e \left(a^2 - b^2 + c^2\right) (a + b \operatorname{Cosh}[d + e x] + c \operatorname{Sinh}[d + e x])}$$

- **Program code:**

```
Int[(A_.+B_.*Cosh[d_.+e_.*x_]+C_.*Sinh[d_.+e_.*x_])/(a_.+b_.*Cosh[d_.+e_.*x_]+c_.*Sinh[d_.+e_.*x_])^
  -(c*B-b*C-(a*C-c*A)*Cosh[d+e*x]+(b*A-a*B)*Sinh[d+e*x])/
    (e*(a^2-b^2+c^2)*(a+b*Cosh[d+e*x]+c*Sinh[d+e*x])) /;
FreeQ[{a,b,c,d,e,A,B,C},x] && NonzeroQ[a^2-b^2+c^2] && ZeroQ[a*A-b*B+c*C]
```

- **Reference: G&R 2.451.1 with $B = 0$, $n = -2$ and $a A + c C = 0$**

```
Int[(A_.+C_.*Sinh[d_.+e_.*x_])/(a_.+b_.*Cosh[d_.+e_.*x_]+c_.*Sinh[d_.+e_.*x_])^2,x_Symbol] :=
  (b*C+(a*C-c*A)*Cosh[d+e*x]-b*A*Sinh[d+e*x])/(e*(a^2-b^2+c^2)*(a+b*Cosh[d+e*x]+c*Sinh[d+e*x])) /;
FreeQ[{a,b,c,d,e,A,C},x] && NonzeroQ[a^2-b^2+c^2] && ZeroQ[a*A+c*C]
```

- **Reference: G&R 2.451.1 with $C = 0$, $n = -2$ and $a A - b B = 0$**

```
Int[(A_.+B_.*Cosh[d_.+e_.*x_])/(a_.+b_.*Cosh[d_.+e_.*x_]+c_.*Sinh[d_.+e_.*x_])^2,x_Symbol] :=
  -(c*B+c*A*Cosh[d+e*x]+(b*A-a*B)*Sinh[d+e*x])/(e*(a^2-b^2+c^2)*(a+b*Cosh[d+e*x]+c*Sinh[d+e*x])) /;
FreeQ[{a,b,c,d,e,A,B},x] && NonzeroQ[a^2-b^2+c^2] && ZeroQ[a*A-b*B]
```

- **Reference:** G&R 2.451.1 with $n = -2$

- **Rule:** If $a^2 - b^2 + c^2 \neq 0 \wedge a A - b B + c C \neq 0$, then

$$\int \frac{A + B \operatorname{Cosh}[d + e x] + C \operatorname{Sinh}[d + e x]}{(a + b \operatorname{Cosh}[d + e x] + c \operatorname{Sinh}[d + e x])^2} \, dx \longrightarrow$$

$$-\frac{c B - b C - (a C - c A) \operatorname{Cosh}[d + e x] + (b A - a B) \operatorname{Sinh}[d + e x]}{e \left(a^2 - b^2 + c^2\right) (a + b \operatorname{Cosh}[d + e x] + c \operatorname{Sinh}[d + e x])} +$$

$$\frac{a A - b B + c C}{a^2 - b^2 + c^2} \int \frac{1}{a + b \operatorname{Cosh}[d + e x] + c \operatorname{Sinh}[d + e x]} \, dx$$

- **Program code:**

```
Int[(A_.+B_.*Cosh[d_.+e_.*x_]+C_.*Sinh[d_.+e_.*x_])/(a_.+b_.*Cosh[d_.+e_.*x_]+c_.*Sinh[d_.+e_.*x_])^
  -(c*B-b*C-(a*C-c*A)*Cosh[d+e*x]+(b*A-a*B)*Sinh[d+e*x])/
     (e*(a^2-b^2+c^2)*(a+b*Cosh[d+e*x]+c*Sinh[d+e*x])) +
  Dist[(a*A-b*B+c*C)/(a^2-b^2+c^2),Int[1/(a+b*Cosh[d+e*x]+c*Sinh[d+e*x]),x]] /;
FreeQ[{a,b,c,d,e,A,B,C},x] && NonzeroQ[a^2-b^2+c^2] && NonzeroQ[a*A-b*B+c*C]
```

- **Reference:** G&R 2.451.1 with $B = 0$ and $n = -2$

```
Int[(A_.+C_.*Sinh[d_.+e_.*x_])/(a_.+b_.*Cosh[d_.+e_.*x_]+c_.*Sinh[d_.+e_.*x_])^2,x_Symbol] :=
  (b*C+(a*C-c*A)*Cosh[d+e*x]-b*A*Sinh[d+e*x])/(e*(a^2-b^2+c^2)*(a+b*Cosh[d+e*x]+c*Sinh[d+e*x])) +
  Dist[(a*A+c*C)/(a^2-b^2+c^2),Int[1/(a+b*Cosh[d+e*x]+c*Sinh[d+e*x]),x]] /;
FreeQ[{a,b,c,d,e,A,C},x] && NonzeroQ[a^2-b^2+c^2] && NonzeroQ[a*A+c*C]
```

- **Reference:** G&R 2.451.1 with $C = 0$ and $n = -2$

```
Int[(A_.+B_.*Cosh[d_.+e_.*x_])/(a_.+b_.*Cosh[d_.+e_.*x_]+c_.*Sinh[d_.+e_.*x_])^2,x_Symbol] :=
  -(c*B+c*A*Cosh[d+e*x]+(b*A-a*B)*Sinh[d+e*x])/(e*(a^2-b^2+c^2)*(a+b*Cosh[d+e*x]+c*Sinh[d+e*x])) +
  Dist[(a*A-b*B)/(a^2-b^2+c^2),Int[1/(a+b*Cosh[d+e*x]+c*Sinh[d+e*x]),x]] /;
FreeQ[{a,b,c,d,e,A,B},x] && NonzeroQ[a^2-b^2+c^2] && NonzeroQ[a*A-b*B]
```

- **Derivation: Algebraic simplification**
- **Basis:** $(A+B z)(a+b z)^n = \frac{B}{b}(a+b z)^{n+1} + \frac{(A b-a B)}{b}(a+b z)^n$
- **Rule: If** $b C - c B = 0 \bigwedge b A - a B \neq 0 \bigwedge \left(n = -\frac{1}{2} \bigvee a^2 - b^2 + c^2 = 0\right)$, **then**

$$\int (A+B\,\text{Cosh}[d+e x]+C\,\text{Sinh}[d+e x])\,(a+b\,\text{Cosh}[d+e x]+c\,\text{Sinh}[d+e x])^n\,dx \longrightarrow$$
$$\frac{B}{b}\int (a+b\,\text{Cosh}[d+e x]+c\,\text{Sinh}[d+e x])^{n+1}\,dx+\frac{b A-a B}{b}\int (a+b\,\text{Cosh}[d+e x]+c\,\text{Sinh}[d+e x])^n\,dx$$

- **Program code:**

```
Int[(A_.+B_.*Cosh[d_.+e_.*x_]+C_.*Sinh[d_.+e_.*x_])*(a_+b_.*Cosh[d_.+e_.*x_]+c_.*Sinh[d_.+e_.*x_])^n
  Dist[B/b,Int[(a+b*Cosh[d+e*x]+c*Sinh[d+e*x])^(n+1),x]] +
  Dist[(b*A-a*B)/b,Int[(a+b*Cosh[d+e*x]+c*Sinh[d+e*x])^n,x]] /;
FreeQ[{a,b,c,d,e,A,B,C},x] && ZeroQ[b*C-c*B] && NonzeroQ[b*A-a*B] && RationalQ[n] && (n==-1/2 || Zer
```

- **Reference: G&R 2.451.1**
- **Rule: If** $a^2 - b^2 + c^2 \neq 0 \bigwedge n < -1 \bigwedge n \neq -2$, **then**

$$\int (A+B\,\text{Cosh}[d+e x]+C\,\text{Sinh}[d+e x])\,(a+b\,\text{Cosh}[d+e x]+c\,\text{Sinh}[d+e x])^n\,dx \longrightarrow$$
$$\frac{(c B-b C-(a C-c A)\,\text{Cosh}[d+e x]+(b A-a B)\,\text{Sinh}[d+e x])\,(a+b\,\text{Cosh}[d+e x]+c\,\text{Sinh}[d+e x])^{n+1}}{e\,(n+1)\left(a^2-b^2+c^2\right)}+$$
$$\frac{1}{(n+1)\left(a^2-b^2+c^2\right)}$$
$$\int ((n+1)\,(a A-b B+c C)-(n+2)\,(b A-a B)\,\text{Cosh}[d+e x]+(n+2)\,(a C-c A)\,\text{Sinh}[d+e x])$$
$$(a+b\,\text{Cosh}[d+e x]+c\,\text{Sinh}[d+e x])^{n+1}\,dx$$

- **Program code:**

```
Int[(A_.+B_.*Cosh[d_.+e_.*x_]+C_.*Sinh[d_.+e_.*x_])*(a_.+b_.*Cosh[d_.+e_.*x_]+c_.*Sinh[d_.+e_.*x_])^
  (c*B-b*C-(a*C-c*A)*Cosh[d+e*x]+(b*A-a*B)*Sinh[d+e*x])*(a+b*Cosh[d+e*x]+c*Sinh[d+e*x])^(n+1)/
    (e*(n+1)*(a^2-b^2+c^2)) +
  Dist[1/((n+1)*(a^2-b^2+c^2)),
    Int[((n+1)*(a*A-b*B+c*C)-(n+2)*(b*A-a*B)*Cosh[d+e*x]+(n+2)*(a*C-c*A)*Sinh[d+e*x])*
      (a+b*Cosh[d+e*x]+c*Sinh[d+e*x])^(n+1),x]] /;
FreeQ[{a,b,c,d,e,A,B,C},x] && NonzeroQ[a^2-b^2+c^2] && RationalQ[n] && n<-1 && n≠-2
```

- **Reference: G&R 2.451.1 with** B = 0

```
Int[(A_.+C_.*Sinh[d_.+e_.*x_])*(a_.+b_.*Cosh[d_.+e_.*x_]+c_.*Sinh[d_.+e_.*x_])^n_,x_Symbol] :=
  -(b*C+(a*C-c*A)*Cosh[d+e*x]-b*A*Sinh[d+e*x])*(a+b*Cosh[d+e*x]+c*Sinh[d+e*x])^(n+1)/
    (e*(n+1)*(a^2-b^2+c^2)) +
  Dist[1/((n+1)*(a^2-b^2+c^2)),
    Int[((n+1)*(a*A+c*C)-(n+2)*b*A*Cosh[d+e*x]+(n+2)*(a*C-c*A)*Sinh[d+e*x])*
      (a+b*Cosh[d+e*x]+c*Sinh[d+e*x])^(n+1),x]] /;
FreeQ[{a,b,c,d,e,A,C},x] && NonzeroQ[a^2-b^2+c^2] && RationalQ[n] && n<-1 && n≠-2
```

- **Reference: G&R 2.451.1 with** C = 0

```
Int[(A_.+B_.*Cosh[d_.+e_.*x_])*(a_.+b_.*Cosh[d_.+e_.*x_]+c_.*Sinh[d_.+e_.*x_])^n_,x_Symbol] :=
  (c*B+c*A*Cosh[d+e*x]+(b*A-a*B)*Sinh[d+e*x])*(a+b*Cosh[d+e*x]+c*Sinh[d+e*x])^(n+1)/
    (e*(n+1)*(a^2-b^2+c^2)) +
  Dist[1/((n+1)*(a^2-b^2+c^2)),
    Int[((n+1)*(a*A-b*B)-(n+2)*(b*A-a*B)*Cosh[d+e*x]-(n+2)*c*A*Sinh[d+e*x])*
      (a+b*Cosh[d+e*x]+c*Sinh[d+e*x])^(n+1),x]] /;
FreeQ[{a,b,c,d,e,A,B},x] && NonzeroQ[a^2-b^2+c^2] && RationalQ[n] && n<-1 && n≠-2
```

- **Reference: G&R 2.451.1 inverted**

- **Rule: If** $a^2 - b^2 + c^2 \neq 0 \wedge n > 0$**, then**

$$\int (A + B\,\text{Cosh}[d+e\,x] + C\,\text{Sinh}[d+e\,x])\,(a + b\,\text{Cosh}[d+e\,x] + c\,\text{Sinh}[d+e\,x])^n\,dx \longrightarrow$$

$$\frac{(-B\,c + b\,C + a\,C\,\text{Cosh}[d+e\,x] + a\,B\,\text{Sinh}[d+e\,x])\,(a + b\,\text{Cosh}[d+e\,x] + c\,\text{Sinh}[d+e\,x])^n}{a\,e\,(n+1)} +$$

$$\frac{1}{a\,(n+1)} \int \left(a\,(b\,B - c\,C)\,n + a^2\,A\,(n+1) + \left(a^2\,B\,n - c\,(b\,C - c\,B)\,n + a\,b\,A\,(n+1)\right)\text{Cosh}[d+e\,x] + \left(a^2\,C\,n - b\,(b\,C - c\,B)\,n + a\,c\,A\,(n+1)\right)\text{Sinh}[d+e\,x]\right) \cdot (a + b\,\text{Cosh}[d+e\,x] + c\,\text{Sinh}[d+e\,x])^{n-1}\,dx$$

- **Program code:**

```
Int[(A_.+B_.*Cosh[d_.+e_.*x_]+C_.*Sinh[d_.+e_.*x_])*(a_+b_.*Cosh[d_.+e_.*x_]+c_.*Sinh[d_.+e_.*x_])^n
  (-B*c+b*C+a*C*Cosh[d+e*x]+a*B*Sinh[d+e*x])*(a+b*Cosh[d+e*x]+c*Sinh[d+e*x])^n/(a*e*(n+1)) +
  Dist[1/(a*(n+1)),
    Int[(a*(b*B-c*C)*n + a^2*A*(n+1) +
        (a^2*B*n - c*(b*C-c*B)*n + a*b*A*(n+1))*Cosh[d+e*x] +
        (a^2*C*n - b*(b*C-c*B)*n + a*c*A*(n+1))*Sinh[d+e*x])*
      (a+b*Cosh[d+e*x]+c*Sinh[d+e*x])^(n-1), x]] /;
FreeQ[{a,b,c,d,e,A,B,C},x] && NonzeroQ[a^2-b^2+c^2] && RationalQ[n] && n>0
```

- **Reference: G&R 2.451.1 inverted with B = 0**

```
Int[(A_.+C_.*Sinh[d_.+e_.*x_])*(a_+b_.*Cosh[d_.+e_.*x_]+c_.*Sinh[d_.+e_.*x_])^n_,x_Symbol] :=
  (b*C+a*C*Cosh[d+e*x])*(a+b*Cosh[d+e*x]+c*Sinh[d+e*x])^n/(a*e*(n+1)) +
  Dist[1/(a*(n+1)),
    Int[(-a*c*C*n+a^2*A*(n+1)-b*(c*C*n-a*A*(n+1))*Cosh[d+e*x]+(a^2*C*n-b^2*C*n+a*c*A*(n+1))*Sinh[d+e
      (a+b*Cosh[d+e*x]+c*Sinh[d+e*x])^(n-1), x]] /;
FreeQ[{a,b,c,d,e,A,C},x] && NonzeroQ[a^2-b^2+c^2] && RationalQ[n] && n>0
```

- **Reference: G&R 2.451.1 inverted with C = 0**

```
Int[(A_.+B_.*Cosh[d_.+e_.*x_])*(a_+b_.*Cosh[d_.+e_.*x_]+c_.*Sinh[d_.+e_.*x_])^n_,x_Symbol] :=
  (-B*c+a*B*Sinh[d+e*x])*(a+b*Cosh[d+e*x]+c*Sinh[d+e*x])^n/(a*e*(n+1)) +
  Dist[1/(a*(n+1)),
    Int[(a*b*B*n+a^2*A*(n+1)+(a^2*B*n+c^2*B*n+a*b*A*(n+1))*Cosh[d+e*x]+c*(b*B*n+a*A*(n+1))*Sinh[d+e*
      (a+b*Cosh[d+e*x]+c*Sinh[d+e*x])^(n-1), x]] /;
FreeQ[{a,b,c,d,e,A,B},x] && NonzeroQ[a^2-b^2+c^2] && RationalQ[n] && n>0
```