$$\int \text{Zeta}[s, a + b\,x]\,dx$$

- **Derivation: Algebraic simplification**
- **Basis:** $\zeta(2, z) = \psi^{(1)}(z)$
- **Rule:**

$$\int \text{Zeta}[2, a + b\,x]\,dx \longrightarrow \int \text{PolyGamma}[1, a + b\,x]\,dx$$

- **Program code:**

```
Int[Zeta[2,a_.+b_.*x_],x_Symbol] :=
  Int[PolyGamma[1,a+b*x],x] /;
FreeQ[{a,b},x]
```

- **Derivation: Primitive rule**
- **Basis:** $\frac{\partial \zeta(s,z)}{\partial z} = -s\,\zeta(s+1, z)$
- **Rule: If** $s \neq 1 \bigwedge s \neq 2$**, then**

$$\int \text{Zeta}[s, a + b\,x]\,dx \longrightarrow -\frac{\text{Zeta}[s-1, a + b\,x]}{b\,(s-1)}$$

- **Program code:**

```
Int[Zeta[s_,a_.+b_.*x_],x_Symbol] :=
  -Zeta[s-1,a+b*x]/(b*(s-1)) /;
FreeQ[{a,b,s},x] && NonzeroQ[s-1] && NonzeroQ[s-2]
```

$$\int x^m \, \text{Zeta}[s, a + b\,x] \, dx$$

- **Derivation: Algebraic simplification**
- **Basis:** $\zeta(2, z) = \psi^{(1)}(z)$
- **Rule: If** $m \in \mathbb{Q}$**, then**

$$\int x^m \, \text{Zeta}[2, a + b\,x] \, dx \;\longrightarrow\; \int x^m \, \text{PolyGamma}[1, a + b\,x] \, dx$$

- **Program code:**

```
Int[x_^m_.*Zeta[2,a_.+b_.*x_],x_Symbol] :=
  Int[x^m*PolyGamma[1,a+b*x],x] /;
FreeQ[{a,b},x] && RationalQ[m]
```

- **Derivation: Integration by parts**
- **Rule: If** $m > 0 \bigwedge s \neq 1 \bigwedge s \neq 2$**, then**

$$\int x^m \, \text{Zeta}[s, a + b\,x] \, dx \;\longrightarrow\; -\frac{x^m \, \text{Zeta}[s - 1, a + b\,x]}{b\,(s - 1)} + \frac{m}{b\,(s - 1)} \int x^{m-1} \, \text{Zeta}[s - 1, a + b\,x] \, dx$$

- **Program code:**

```
Int[x_^m_.*Zeta[s_,a_.+b_.*x_],x_Symbol] :=
  -x^m*Zeta[s-1,a+b*x]/(b*(s-1)) +
  Dist[m/(b*(s-1)),Int[x^(m-1)*Zeta[s-1,a+b*x],x]] /;
FreeQ[{a,b,s},x] && RationalQ[m] && m>0 && NonzeroQ[s-1] && NonzeroQ[s-2]
```

- **Derivation: Inverted integration by parts**
- **Rule: If** $m < -1 \bigwedge s \neq 1 \bigwedge s \neq 2$**, then**

$$\int x^m \, \text{Zeta}[s, a + b\,x] \, dx \;\longrightarrow\; \frac{x^{m+1} \, \text{Zeta}[s, a + b\,x]}{m + 1} + \frac{b\,s}{m + 1} \int x^{m+1} \, \text{Zeta}[s + 1, a + b\,x] \, dx$$

- **Program code:**

```
Int[x_^m_.*Zeta[s_,a_.+b_.*x_],x_Symbol] :=
  x^(m+1)*Zeta[s,a+b*x]/(m+1) +
  Dist[b*s/(m+1),Int[x^(m+1)*Zeta[s+1,a+b*x],x]] /;
FreeQ[{a,b,s},x] && RationalQ[m] && m<-1 && NonzeroQ[s-1] && NonzeroQ[s-2]
```