$$\int \frac{x\,(A+B\,\mathrm{Sinh}[c+d\,x])}{(a+b\,\mathrm{Sinh}[c+d\,x])^2}\,dx$$

- **Derivation: Integration by parts**
- **Rule: If** $a\,A+b\,B=0$**, then**

$$\int \frac{x\,(A+B\,\mathrm{Sinh}[c+d\,x])}{(a+b\,\mathrm{Sinh}[c+d\,x])^2}\,dx \longrightarrow \frac{B\,x\,\mathrm{Cosh}[c+d\,x]}{a\,d\,(a+b\,\mathrm{Sinh}[c+d\,x])}-\frac{B}{a\,d}\int \frac{\mathrm{Cosh}[c+d\,x]}{a+b\,\mathrm{Sinh}[c+d\,x]}\,dx$$

- **Program code:**

```
Int[x_*(A_+B_.*Sinh[c_.+d_.*x_])/(a_+b_.*Sinh[c_.+d_.*x_])^2,x_Symbol] :=
  B*x*Cosh[c+d*x]/(a*d*(a+b*Sinh[c+d*x])) -
  Dist[B/(a*d),Int[Cosh[c+d*x]/(a+b*Sinh[c+d*x]),x]] /;
FreeQ[{a,b,c,d,A,B},x] && ZeroQ[a*A+b*B]
```

```
Int[x_*(A_+B_.*Cosh[c_.+d_.*x_])/(a_+b_.*Cosh[c_.+d_.*x_])^2,x_Symbol] :=
  B*x*Sinh[c+d*x]/(a*d*(a+b*Cosh[c+d*x])) -
  Dist[B/(a*d),Int[Sinh[c+d*x]/(a+b*Cosh[c+d*x]),x]] /;
FreeQ[{a,b,c,d,A,B},x] && ZeroQ[a*A-b*B]
```

$$\int \mathrm{Sinh}[a+b\,x]^m\,\mathrm{Tanh}[a+b\,x]^n\,dx$$

- **Reference: G&R 2.423.18'**
- **Derivation: Algebraic expansion**
- **Basis:** $\mathrm{Sinh}[z]\,\mathrm{Tanh}[z] = \mathrm{Cosh}[z] - \mathrm{Sech}[z]$
- **Rule:**

$$\int \mathrm{Sinh}[a+b\,x]\,\mathrm{Tanh}[a+b\,x]\,dx \longrightarrow \frac{\mathrm{Sinh}[a+b\,x]}{b} - \int \mathrm{Sech}[a+b\,x]\,dx$$

- **Program code:**

```
Int[Sinh[a_.+b_.*x_]*Tanh[a_.+b_.*x_],x_Symbol] :=
  Sinh[a+b*x]/b - Int[Sech[a+b*x],x] /;
FreeQ[{a,b},x]
```

- **Reference: G&R 2.423.34'**

```
Int[Cosh[a_.+b_.*x_]*Coth[a_.+b_.*x_],x_Symbol] :=
  Cosh[a+b*x]/b + Int[Csch[a+b*x],x] /;
FreeQ[{a,b},x]
```

- **Rule: If** $m+n-1=0$, **then**

$$\int \mathrm{Sinh}[a+b\,x]^m\,\mathrm{Tanh}[a+b\,x]^n\,dx \longrightarrow \frac{\mathrm{Sinh}[a+b\,x]^m\,\mathrm{Tanh}[a+b\,x]^{n-1}}{b\,m}$$

- **Program code:**

```
Int[Sinh[a_.+b_.*x_]^m_*Tanh[a_.+b_.*x_]^n_,x_Symbol]:=
  Sinh[a+b*x]^m*Tanh[a+b*x]^(n-1)/(b*m) /;
FreeQ[{a,b,m,n},x] && ZeroQ[m+n-1]
```

```
Int[Cosh[a_.+b_.*x_]^m_*Coth[a_.+b_.*x_]^n_,x_Symbol] :=
  Cosh[a+b*x]^m*Coth[a+b*x]^(n-1)/(b*m) /;
FreeQ[{a,b,m,n},x] && ZeroQ[m+n-1]
```

- **Derivation: Integration by substitution**

- **Basis: If** $m, n, \frac{m+n-1}{2} \in \mathbb{Z}$**, then** $\text{Sinh}[z]^m \, \text{Tanh}[z]^n = \frac{\left(-1+\text{Cosh}[z]^2\right)^{\frac{m+n-1}{2}}}{\text{Cosh}[z]^n} \, \partial_z \text{Cosh}[z]$

- **Note: This rule is used if** $m + n$ **is odd since it requires fewer steps and results in a simpler antiderivative than the other rules in this section.**

- **Rule: If** $m, n, \frac{m+n-1}{2} \in \mathbb{Z}$**, then**

$$\int \text{Sinh}[a+b\,x]^m \, \text{Tanh}[a+b\,x]^n \, dx \longrightarrow \frac{1}{b} \, \text{Subst}\left[\int \frac{\left(-1+x^2\right)^{\frac{m+n-1}{2}}}{x^n} \, dx, \, x, \, \text{Cosh}[a+b\,x]\right]$$

- **Program code:**

```
Int[Sinh[a_.+b_.*x_]^m_.*Tanh[a_.+b_.*x_]^n_.,x_Symbol] :=
  Dist[1/b,Subst[Int[Regularize[(-1+x^2)^((m+n-1)/2)/x^n,x],x],x,Cosh[a+b*x]]] /;
FreeQ[{a,b},x] && IntegersQ[m,n,(m+n-1)/2]
```

- **Basis: If** $m, n, \frac{m+n-1}{2} \in \mathbb{Z}$**, then** $\text{Cosh}[z]^m \, \text{Coth}[z]^n = \frac{\left(1+\text{Sinh}[z]^2\right)^{\frac{m+n-1}{2}}}{\text{Sinh}[z]^n} \, \partial_z \text{Sinh}[z]$

```
Int[Cosh[a_.+b_.*x_]^m_.*Coth[a_.+b_.*x_]^n_.,x_Symbol] :=
  Dist[1/b,Subst[Int[Regularize[(1+x^2)^((m+n-1)/2)/x^n,x],x],x,Sinh[a+b*x]]] /;
FreeQ[{a,b},x] && IntegersQ[m,n,(m+n-1)/2]
```

- **Reference: G&R 2.411.1, CRC 567a**

- **Rule: If** $m > 1 \land n < -1$**, then**

$$\int \text{Sinh}[a+b\,x]^m \, \text{Tanh}[a+b\,x]^n \, dx \longrightarrow \frac{\text{Sinh}[a+b\,x]^m \, \text{Tanh}[a+b\,x]^{n+1}}{b\,m} - \frac{n+1}{m} \int \text{Sinh}[a+b\,x]^{m-2} \, \text{Tanh}[a+b\,x]^{n+2} \, dx$$

- **Program code:**

```
Int[Sinh[a_.+b_.*x_]^m_*Tanh[a_.+b_.*x_]^n_,x_Symbol] :=
  Sinh[a+b*x]^m*Tanh[a+b*x]^(n+1)/(b*m) -
  Dist[(n+1)/m,Int[Sinh[a+b*x]^(m-2)*Tanh[a+b*x]^(n+2),x]] /;
FreeQ[{a,b},x] && RationalQ[{m,n}] && m>1 && n<-1
```

- **Reference: G&R 2.411.2, CRC 567b**

```
Int[Cosh[a_.+b_.*x_]^m_*Coth[a_.+b_.*x_]^n_,x_Symbol] :=
  Cosh[a+b*x]^m*Coth[a+b*x]^(n+1)/(b*m) +
  Dist[(n+1)/m,Int[Cosh[a+b*x]^(m-2)*Coth[a+b*x]^(n+2),x]] /;
FreeQ[{a,b},x] && RationalQ[{m,n}] && m>1 && n<-1
```

- **Reference: G&R 2.411.6, CRC 568b**

- **Rule: If** $m < -1 \wedge n > 1$**, then**

$$\int \text{Sinh}[a+b\,x]^m\,\text{Tanh}[a+b\,x]^n\,dx \longrightarrow \frac{\text{Sinh}[a+b\,x]^{m+2}\,\text{Tanh}[a+b\,x]^{n-1}}{b\,(n-1)} - \frac{m+2}{n-1}\int \text{Sinh}[a+b\,x]^{m+2}\,\text{Tanh}[a+b\,x]^{n-2}\,dx$$

- **Program code:**

```
Int[Sinh[a_.+b_.*x_]^m_*Tanh[a_.+b_.*x_]^n_,x_Symbol] :=
  Sinh[a+b*x]^(m+2)*Tanh[a+b*x]^(n-1)/(b*(n-1)) -
  Dist[(m+2)/(n-1),Int[Sinh[a+b*x]^(m+2)*Tanh[a+b*x]^(n-2),x]] /;
FreeQ[{a,b},x] && RationalQ[{m,n}] && m<-1 && n>1
```

- **Reference: G&R 2.411.5, CRC 568a**

```
Int[Cosh[a_.+b_.*x_]^m_*Coth[a_.+b_.*x_]^n_,x_Symbol] :=
  -Cosh[a+b*x]^(m+2)*Coth[a+b*x]^(n-1)/(b*(n-1)) +
  Dist[(m+2)/(n-1),Int[Cosh[a+b*x]^(m+2)*Coth[a+b*x]^(n-2),x]] /;
FreeQ[{a,b},x] && RationalQ[{m,n}] && m<-1 && n>1
```

- **Reference: G&R 2.411.2, CRC 567b**

- **Rule: If** $m > 1$**, then**

$$\int \text{Sinh}[a+b\,x]^m\,\text{Tanh}[a+b\,x]^n\,dx \longrightarrow \frac{\text{Sinh}[a+b\,x]^{m}\,\text{Tanh}[a+b\,x]^{n-1}}{b\,m} - \frac{m+n-1}{m}\int \text{Sinh}[a+b\,x]^{m-2}\,\text{Tanh}[a+b\,x]^{n}\,dx$$

- **Program code:**

```
Int[Sinh[a_.+b_.*x_]^m_*Tanh[a_.+b_.*x_]^n_.,x_Symbol]:=
  Sinh[a+b*x]^m*Tanh[a+b*x]^(n-1)/(b*m) -
  Dist[(m+n-1)/m,Int[Sinh[a+b*x]^(m-2)*Tanh[a+b*x]^n,x]] /;
FreeQ[{a,b,n},x] && RationalQ[m] && m>1
```

- **Reference: G&R 2.411.1, CRC 567a**

```
Int[Cosh[a_.+b_.*x_]^m_*Coth[a_.+b_.*x_]^n_.,x_Symbol] :=
  Cosh[a+b*x]^m*Coth[a+b*x]^(n-1)/(b*m) +
  Dist[(m+n-1)/m,Int[Cosh[a+b*x]^(m-2)*Coth[a+b*x]^n,x]] /;
FreeQ[{a,b,n},x] && RationalQ[m] && m>1
```

- **Reference: G&R 2.411.3**

- **Rule: If $n > 1$, then**

$$\int \text{Sinh}[a+b\,x]^m\,\text{Tanh}[a+b\,x]^n\,dx \longrightarrow -\frac{\text{Sinh}[a+b\,x]^m\,\text{Tanh}[a+b\,x]^{n-1}}{b\,(n-1)} + \frac{m+n-1}{n-1}\int \text{Sinh}[a+b\,x]^m\,\text{Tanh}[a+b\,x]^{n-2}\,dx$$

- **Program code:**

```
Int[Sinh[a_.+b_.*x_]^m_.*Tanh[a_.+b_.*x_]^n_,x_Symbol] :=
  -Sinh[a+b*x]^m*Tanh[a+b*x]^(n-1)/(b*(n-1)) +
  Dist[(m+n-1)/(n-1),Int[Sinh[a+b*x]^m*Tanh[a+b*x]^(n-2),x]] /;
FreeQ[{a,b,m},x] && RationalQ[n] && n>1
```

- **Reference: G&R 2.411.4**

```
Int[Cosh[a_.+b_.*x_]^m_.*Coth[a_.+b_.*x_]^n_,x_Symbol] :=
  -Cosh[a+b*x]^m*Coth[a+b*x]^(n-1)/(b*(n-1)) +
  Dist[(m+n-1)/(n-1),Int[Cosh[a+b*x]^m*Coth[a+b*x]^(n-2),x]] /;
FreeQ[{a,b,m},x] && RationalQ[n] && n>1
```

- **Reference: G&R 2.411.5, CRC 568a**

- **Rule: If $m < -1 \land m+n+1 \neq 0$, then**

$$\int \text{Sinh}[a+b\,x]^m\,\text{Tanh}[a+b\,x]^n\,dx \longrightarrow \frac{\text{Sinh}[a+b\,x]^{m+2}\,\text{Tanh}[a+b\,x]^{n-1}}{b\,(m+n+1)} - \frac{m+2}{m+n+1}\int \text{Sinh}[a+b\,x]^{m+2}\,\text{Tanh}[a+b\,x]^n\,dx$$

- **Program code:**

```
Int[Sinh[a_.+b_.*x_]^m_*Tanh[a_.+b_.*x_]^n_.,x_Symbol]:=
  Sinh[a+b*x]^(m+2)*Tanh[a+b*x]^(n-1)/(b*(m+n+1)) -
  Dist[(m+2)/(m+n+1),Int[Sinh[a+b*x]^(m+2)*Tanh[a+b*x]^n,x]] /;
FreeQ[{a,b,n},x] && RationalQ[m] && m<-1 && NonzeroQ[m+n+1]
```

- **Reference: G&R 2.411.6, CRC 568b**

```
Int[Cosh[a_.+b_.*x_]^m_*Coth[a_.+b_.*x_]^n_.,x_Symbol] :=
  -Cosh[a+b*x]^(m+2)*Coth[a+b*x]^(n-1)/(b*(m+n+1)) +
  Dist[(m+2)/(m+n+1),Int[Cosh[a+b*x]^(m+2)*Coth[a+b*x]^n,x]] /;
FreeQ[{a,b,n},x] && RationalQ[m] && m<-1 && NonzeroQ[m+n+1]
```

- **Reference: G&R 2.411.4**

- **Rule: If** $n < -1 \land m + n + 1 \neq 0$**, then**

$$\int \text{Sinh}[a + b\,x]^m\,\text{Tanh}[a + b\,x]^n\,dx \longrightarrow \frac{\text{Sinh}[a + b\,x]^m\,\text{Tanh}[a + b\,x]^{n+1}}{b\,(m + n + 1)} + \frac{n + 1}{m + n + 1}\int \text{Sinh}[a + b\,x]^m\,\text{Tanh}[a + b\,x]^{n+2}\,dx$$

- **Program code:**

```
Int[Sinh[a_.+b_.*x_]^m_.*Tanh[a_.+b_.*x_]^n_,x_Symbol]:=
  Sinh[a+b*x]^m*Tanh[a+b*x]^(n+1)/(b*(m+n+1)) +
  Dist[(n+1)/(m+n+1),Int[Sinh[a+b*x]^m*Tanh[a+b*x]^(n+2),x]] /;
FreeQ[{a,b,m},x] && RationalQ[n] && n<-1 && NonzeroQ[m+n+1]
```

- **Reference: G&R 2.411.3**

```
Int[Cosh[a_.+b_.*x_]^m_.*Coth[a_.+b_.*x_]^n_,x_Symbol] :=
  Cosh[a+b*x]^m*Coth[a+b*x]^(n+1)/(b*(m+n+1)) +
  Dist[(n+1)/(m+n+1),Int[Cosh[a+b*x]^m*Coth[a+b*x]^(n+2),x]] /;
FreeQ[{a,b,m},x] && RationalQ[n] && n<-1 && NonzeroQ[m+n+1]
```

$\int u\,\text{Sinh}[v]\,\text{Hyper}[w]\,dx$

- **Derivation: Algebraic expansion**
- **Basis:** $\text{Sinh}[v]\,\text{Cosh}[w] = \frac{1}{2}\,\text{Sinh}[v+w] + \frac{1}{2}\,\text{Sinh}[v-w]$
- **Rule: If** $v,\ w \in \mathbb{P}x\ \wedge\ v+w \neq 0\ \wedge\ v-w \neq 0$**, then**

$$\int u\,\text{Sinh}[v]\,\text{Cosh}[w]\,dx \longrightarrow \frac{1}{2}\int u\,\text{Sinh}[v+w]\,dx + \frac{1}{2}\int u\,\text{Sinh}[v-w]\,dx$$

- **Program code:**

```
Int[u_.*Sinh[v_]*Cosh[w_],x_Symbol] :=
  Dist[1/2,Int[u*Regularize[Sinh[v+w],x],x]] +
  Dist[1/2,Int[u*Regularize[Sinh[v-w],x],x]] /;
(PolynomialQ[v,x] && PolynomialQ[w,x] || IndependentQ[Cancel[v/w],x]) && NonzeroQ[v+w] && NonzeroQ[v-
```

- **Derivation: Algebraic expansion**
- **Basis:** $\text{Sinh}[v]\,\text{Sinh}[w] = \frac{1}{2}\,\text{Cosh}[v+w] - \frac{1}{2}\,\text{Cosh}[v-w]$
- **Rule: If** $v,\ w \in \mathbb{P}x\ \wedge\ v+w \neq 0\ \wedge\ v-w \neq 0$**, then**

$$\int u\,\text{Sinh}[v]\,\text{Sinh}[w]\,dx \longrightarrow \frac{1}{2}\int u\,\text{Cosh}[v+w]\,dx - \frac{1}{2}\int u\,\text{Cosh}[v-w]\,dx$$

- **Program code:**

```
Int[u_.*Sinh[v_]*Sinh[w_],x_Symbol] :=
  Dist[1/2,Int[u*Regularize[Cosh[v+w],x],x]] -
  Dist[1/2,Int[u*Regularize[Cosh[v-w],x],x]] /;
(PolynomialQ[v,x] && PolynomialQ[w,x] || IndependentQ[Cancel[v/w],x]) && NonzeroQ[v+w] && NonzeroQ[v-
```

- **Basis:** $\text{Cosh}[v]\,\text{Cosh}[w] = \frac{1}{2}\,\text{Cosh}[v-w] + \frac{1}{2}\,\text{Cosh}[v+w]$

```
Int[u_.*Cosh[v_]*Cosh[w_],x_Symbol] :=
  Dist[1/2,Int[u*Regularize[Cosh[v-w],x],x]] +
  Dist[1/2,Int[u*Regularize[Cosh[v+w],x],x]] /;
(PolynomialQ[v,x] && PolynomialQ[w,x] || IndependentQ[Cancel[v/w],x]) && NonzeroQ[v+w] && NonzeroQ[v-
```

- **Derivation: Algebraic expansion**
- **Basis:** $\text{Sinh}[v]\ \text{Tanh}[w] = \text{Cosh}[v] - \text{Cosh}[v-w]\ \text{Sech}[w]$
- **Rule: If** $n>0 \land x \notin v-w \neq 0$**, then**

$$\int u\ \text{Sinh}[v]\ \text{Tanh}[w]^n\, dx \longrightarrow \int u\ \text{Cosh}[v]\ \text{Tanh}[w]^{n-1}\, dx - \text{Cosh}[v-w] \int u\ \text{Sech}[w]\ \text{Tanh}[w]^{n-1}\, dx$$

- **Program code:**

```
Int[u_.*Sinh[v_]*Tanh[w_]^n_.,x_Symbol] :=
  Int[u*Cosh[v]*Tanh[w]^(n-1),x] - Cosh[v-w]*Int[u*Sech[w]*Tanh[w]^(n-1),x] /;
RationalQ[n] && n>0 && FreeQ[v-w,x] && NonzeroQ[v-w]
```

- **Basis:** $\text{Cosh}[v]\ \text{Coth}[w] = \text{Sinh}[v] + \text{Cosh}[v-w]\ \text{Csch}[w]$

```
Int[u_.*Cosh[v_]*Coth[w_]^n_.,x_Symbol] :=
  Int[u*Sinh[v]*Coth[w]^(n-1),x] + Cosh[v-w]*Int[u*Csch[w]*Coth[w]^(n-1),x] /;
RationalQ[n] && n>0 && FreeQ[v-w,x] && NonzeroQ[v-w]
```

- **Derivation: Algebraic expansion**
- **Basis:** $\text{Sinh}[v]\ \text{Coth}[w] = \text{Cosh}[v] + \text{Sinh}[v-w]\ \text{Csch}[w]$
- **Rule: If** $n>0 \land x \notin v-w \neq 0$**, then**

$$\int u\ \text{Sinh}[v]\ \text{Coth}[w]^n\, dx \longrightarrow \int u\ \text{Cosh}[v]\ \text{Coth}[w]^{n-1}\, dx + \text{Sinh}[v-w] \int u\ \text{Csch}[w]\ \text{Coth}[w]^{n-1}\, dx$$

- **Program code:**

```
Int[u_.*Sinh[v_]*Coth[w_]^n_.,x_Symbol] :=
  Int[u*Cosh[v]*Coth[w]^(n-1),x] + Sinh[v-w]*Int[u*Csch[w]*Coth[w]^(n-1),x] /;
RationalQ[n] && n>0 && FreeQ[v-w,x] && NonzeroQ[v-w]
```

- **Basis:** $\text{Cosh}[v]\ \text{Tanh}[w] = \text{Sinh}[v] - \text{Sinh}[v-w]\ \text{Sech}[w]$

```
Int[u_.*Cosh[v_]*Tanh[w_]^n_.,x_Symbol] :=
  Int[u*Sinh[v]*Tanh[w]^(n-1),x] - Sinh[v-w]*Int[u*Sech[w]*Tanh[w]^(n-1),x] /;
RationalQ[n] && n>0 && FreeQ[v-w,x] && NonzeroQ[v-w]
```

- **Derivation: Algebraic expansion**
- **Basis:** $\text{Sinh}[v]\ \text{Sech}[w] = \text{Cosh}[v-w]\ \text{Tanh}[w] + \text{Sinh}[v-w]$
- **Rule: If** $n>0 \land x \notin v-w \neq 0$**, then**

$$\int u\ \text{Sinh}[v]\ \text{Sech}[w]^n\, dx \longrightarrow \text{Cosh}[v-w] \int u\ \text{Tanh}[w]\ \text{Sech}[w]^{n-1}\, dx + \text{Sinh}[v-w] \int u\ \text{Sech}[w]^{n-1}\, dx$$

- **Program code:**

```
Int[u_.*Sinh[v_]*Sech[w_]^n_.,x_Symbol] :=
  Cosh[v-w]*Int[u*Tanh[w]*Sech[w]^(n-1),x] + Sinh[v-w]*Int[u*Sech[w]^(n-1),x] /;
RationalQ[n] && n>0 && FreeQ[v-w,x] && NonzeroQ[v-w]
```

- **Basis:** $\text{Cosh}[v] * \text{Csch}[w] = \text{Cosh}[v-w] * \text{Coth}[w] + \text{Sinh}[v-w]$

```
Int[u_.*Cosh[v_]*Csch[w_]^n_.,x_Symbol] :=
  Cosh[v-w]*Int[u*Coth[w]*Csch[w]^(n-1),x] + Sinh[v-w]*Int[u*Csch[w]^(n-1),x] /;
RationalQ[n] && n>0 && FreeQ[v-w,x] && NonzeroQ[v-w]
```

- **Derivation: Algebraic expansion**
- **Basis:** $\text{Sinh}[v]\ \text{Csch}[w] = \text{Sinh}[v-w]\ \text{Coth}[w] + \text{Cosh}[v-w]$
- **Rule: If** $n>0 \land x \notin v-w \neq 0$**, then**

$$\int u\ \text{Sinh}[v]\ \text{Csch}[w]^n\, dx \longrightarrow \text{Sinh}[v-w] \int u\ \text{Coth}[w]\ \text{Csch}[w]^{n-1}\, dx + \text{Cosh}[v-w] \int u\ \text{Csch}[w]^{n-1}\, dx$$

- **Program code:**

```
Int[u_.*Sinh[v_]*Csch[w_]^n_.,x_Symbol] :=
  Sinh[v-w]*Int[u*Coth[w]*Csch[w]^(n-1),x] + Cosh[v-w]*Int[u*Csch[w]^(n-1),x] /;
RationalQ[n] && n>0 && FreeQ[v-w,x] && NonzeroQ[v-w]
```

- **Basis:** $\text{Cosh}[v]\ \text{Sech}[w] = \text{Sinh}[v-w]\ \text{Tanh}[w] + \text{Cosh}[v-w]$

```
Int[u_.*Cosh[v_]*Sech[w_]^n_.,x_Symbol] :=
  Sinh[v-w]*Int[u*Tanh[w]*Sech[w]^(n-1),x] + Cosh[v-w]*Int[u*Sech[w]^(n-1),x] /;
RationalQ[n] && n>0 && FreeQ[v-w,x] && NonzeroQ[v-w]
```

$$\int x^m \operatorname{Sinh}[a+b\,x^n]^p \operatorname{Cosh}[a+b\,x^n]\,dx$$

- **Reference: G&R 2.479.6**

- **Rule: If** $m, n, p \in \mathbb{Z} \land p \neq -1 \land 0 < n \leq m$, **then**

$$\int x^m \operatorname{Sinh}[a+b\,x^n]^p \operatorname{Cosh}[a+b\,x^n]\,dx \longrightarrow \frac{x^{m-n+1} \operatorname{Sinh}[a+b\,x^n]^{p+1}}{b\,n\,(p+1)} - \frac{m-n+1}{b\,n\,(p+1)} \int x^{m-n} \operatorname{Sinh}[a+b\,x^n]^{p+1}\,dx$$

- **Program code:**

```
Int[x_^m_.*Sinh[a_.+b_.*x_^n_.]^p_.*Cosh[a_.+b_.*x_^n_.],x_Symbol] :=
  x^(m-n+1)*Sinh[a+b*x^n]^(p+1)/(b*n*(p+1)) -
  Dist[(m-n+1)/(b*n*(p+1)),Int[x^(m-n)*Sinh[a+b*x^n]^(p+1),x]] /;
FreeQ[{a,b},x] && IntegersQ[m,n,p] && p≠-1 && 0<n≤m
```

- **Reference: G&R 2.479.3**

```
Int[x_^m_.*Cosh[a_.+b_.*x_^n_.]^p_.*Sinh[a_.+b_.*x_^n_.],x_Symbol] :=
  x^(m-n+1)*Cosh[a+b*x^n]^(p+1)/(b*n*(p+1)) -
  Dist[(m-n+1)/(b*n*(p+1)),Int[x^(m-n)*Cosh[a+b*x^n]^(p+1),x]] /;
FreeQ[{a,b},x] && IntegersQ[m,n,p] && p≠-1 && 0<n≤m
```

$$\int \text{Sinh}[a+b\,x]^m\,\text{Cosh}[a+b\,x]^n\,dx$$

- **Reference: G&R 2.411.5, CRC 568a, A&S 4.5.86a with $m+n+2=0$**
- **Rule: If $m+n+2=0 \wedge m+1\neq 0$, then**

$$\int \text{Sinh}[a+b\,x]^m\,\text{Cosh}[a+b\,x]^n\,dx \longrightarrow \frac{\text{Sinh}[a+b\,x]^{m+1}\,\text{Cosh}[a+b\,x]^{n+1}}{b\,(m+1)}$$

- **Program code:**

```
Int[Sinh[a_.+b_.*x_]^m_.*Cosh[a_.+b_.*x_]^n_,x_Symbol] :=
  Sinh[a+b*x]^(m+1)*Cosh[a+b*x]^(n+1)/(b*(m+1)) /;
FreeQ[{a,b,m,n},x] && ZeroQ[m+n+2] && NonzeroQ[m+1]
```

- **Derivation: Integration by substitution**
- **Basis: If $\frac{n-1}{2}\in\mathbb{Z}$, then $\text{Sinh}[z]^m\,\text{Cosh}[z]^n = \text{Sinh}[z]^m\left(1+\text{Sinh}[z]^2\right)^{\frac{n-1}{2}}\text{Sinh}'[z]$**
- **Note: This rule is used for odd n since it requires fewer steps and results in a simpler antiderivative than the other rules in this section.**
- **Rule: If $\frac{n-1}{2}\in\mathbb{Z} \bigwedge \neg\left(\frac{m-1}{2}\in\mathbb{Z} \bigwedge 0<m<n\right)$, then**

$$\int \text{Sinh}[a+b\,x]^m\,\text{Cosh}[a+b\,x]^n\,dx \longrightarrow \frac{1}{b}\,\text{Subst}\left[\int x^m\left(1+x^2\right)^{\frac{n-1}{2}}dx,\,x,\,\text{Sinh}[a+b\,x]\right]$$

- **Program code:**

```
Int[Sinh[a_.+b_.*x_]^m_*Cosh[a_.+b_.*x_]^n_,x_Symbol] :=
  Dist[1/b,Subst[Int[Regularize[x^m*(1+x^2)^((n-1)/2),x],x],x,Sinh[a+b*x]]] /;
FreeQ[{a,b,m},x] && OddQ[n] && Not[OddQ[m] && 0<m<n]
```

- **Basis: If $\frac{m-1}{2}\in\mathbb{Z}$, then $\text{Sinh}[z]^m\,\text{Cosh}[z]^n = \text{Cosh}[z]^n\left(-1+\text{Cosh}[z]^2\right)^{\frac{m-1}{2}}\text{Cosh}'[z]$**

```
Int[Sinh[a_.+b_.*x_]^m_*Cosh[a_.+b_.*x_]^n_,x_Symbol] :=
  Dist[1/b,Subst[Int[Regularize[x^n*(-1+x^2)^((m-1)/2),x],x],x,Cosh[a+b*x]]] /;
FreeQ[{a,b,n},x] && OddQ[m] && Not[OddQ[n] && 0<n<=m]
```

- **Reference: G&R 2.411.3**
- **Rule: If** $m > 1 \wedge n < -1$**, then**

$$\int \text{Sinh}[a+b\,x]^m \,\text{Cosh}[a+b\,x]^n\,dx \longrightarrow \frac{\text{Sinh}[a+b\,x]^{m-1}\,\text{Cosh}[a+b\,x]^{n+1}}{b\,(n+1)} - \frac{m-1}{n+1}\int \text{Sinh}[a+b\,x]^{m-2}\,\text{Cosh}[a+b\,x]^{n+2}\,dx$$

- **Program code:**

```
Int[Sinh[a_.+b_.*x_]^m_*Cosh[a_.+b_.*x_]^n_,x_Symbol] :=
  Sinh[a+b*x]^(m-1)*Cosh[a+b*x]^(n+1)/(b*(n+1)) -
  Dist[(m-1)/(n+1),Int[Sinh[a+b*x]^(m-2)*Cosh[a+b*x]^(n+2),x]] /;
FreeQ[{a,b},x] && RationalQ[{m,n}] && m>1 && n<-1
```

- **Reference: G&R 2.411.4**

```
Int[Sinh[a_.+b_.*x_]^m_*Cosh[a_.+b_.*x_]^n_,x_Symbol] :=
  Sinh[a+b*x]^(m+1)*Cosh[a+b*x]^(n-1)/(b*(m+1)) -
  Dist[(n-1)/(m+1),Int[Sinh[a+b*x]^(m+2)*Cosh[a+b*x]^(n-2),x]] /;
FreeQ[{a,b},x] && RationalQ[{m,n}] && m<-1 && n>1
```

- **Reference: G&R 2.411.2, CRC 567b, A&S 4.5.85b**
- **Rule: If** $m > 1 \bigwedge \frac{m-1}{2} \notin \mathbb{Z} \bigwedge m+n \neq 0 \bigwedge \neg\left(\frac{n-1}{2} \in \mathbb{Z} \bigwedge n > 1\right)$**, then**

$$\int \text{Sinh}[a+b\,x]^m \,\text{Cosh}[a+b\,x]^n\,dx \longrightarrow \frac{\text{Sinh}[a+b\,x]^{m-1}\,\text{Cosh}[a+b\,x]^{n+1}}{b\,(m+n)} - \frac{m-1}{m+n}\int \text{Sinh}[a+b\,x]^{m-2}\,\text{Cosh}[a+b\,x]^{n}\,dx$$

- **Program code:**

```
Int[Sinh[a_.+b_.*x_]^m_*Cosh[a_.+b_.*x_]^n_,x_Symbol] :=
  Sinh[a+b*x]^(m-1)*Cosh[a+b*x]^(n+1)/(b*(m+n)) -
  Dist[(m-1)/(m+n),Int[Sinh[a+b*x]^(m-2)*Cosh[a+b*x]^n,x]] /;
FreeQ[{a,b,n},x] && RationalQ[m] && m>1 && NonzeroQ[m+n]
```

- **Reference: G&R 2.411.1, CRC 567a, A&S 4.5.85a**

```
Int[Sinh[a_.+b_.*x_]^m_*Cosh[a_.+b_.*x_]^n_,x_Symbol] :=
  Sinh[a+b*x]^(m+1)*Cosh[a+b*x]^(n-1)/(b*(m+n)) +
  Dist[(n-1)/(m+n),Int[Sinh[a+b*x]^m*Cosh[a+b*x]^(n-2),x]] /;
FreeQ[{a,b,m},x] && RationalQ[n] && n>1 && NonzeroQ[m+n]
```

- **Reference: G&R 2.411.5, CRC 568a, A&S 4.5.86a**
- **Rule: If $m < -1 \land m + n + 2 \neq 0$, then**

$$\int \mathrm{Sinh}[a + b\,x]^m \, \mathrm{Cosh}[a + b\,x]^n \, dx \longrightarrow \frac{\mathrm{Sinh}[a + b\,x]^{m+1} \, \mathrm{Cosh}[a + b\,x]^{n+1}}{b\,(m + 1)} - \frac{m + n + 2}{m + 1} \int \mathrm{Sinh}[a + b\,x]^{m+2} \, \mathrm{Cosh}[a + b\,x]^n \, dx$$

- **Program code:**

```
Int[Sinh[a_.+b_.*x_]^m_*Cosh[a_.+b_.*x_]^n_,x_Symbol] :=
  Sinh[a+b*x]^(m+1)*Cosh[a+b*x]^(n+1)/(b*(m+1)) -
  Dist[(m+n+2)/(m+1),Int[Sinh[a+b*x]^(m+2)*Cosh[a+b*x]^n,x]] /;
FreeQ[{a,b,n},x] && RationalQ[m] && m<-1 && NonzeroQ[m+n+2]
```

- **Reference: G&R 2.411.6, CRC 568b, A&S 4.5.86b**

```
Int[Sinh[a_.+b_.*x_]^m_*Cosh[a_.+b_.*x_]^n_,x_Symbol] :=
  -Sinh[a+b*x]^(m+1)*Cosh[a+b*x]^(n+1)/(b*(n+1)) +
  Dist[(m+n+2)/(n+1),Int[Sinh[a+b*x]^m*Cosh[a+b*x]^(n+2),x]] /;
FreeQ[{a,b,m},x] && RationalQ[n] && n<-1 && NonzeroQ[m+n+2]
```

- **Derivation: Integration by substitution**
- **Basis: If $\frac{1}{m} \in \mathbb{Z}$, then** $\frac{\mathrm{Sinh}[z]^m}{\mathrm{Cosh}[z]^m} = \frac{\left(\frac{\mathrm{Sinh}[z]^m}{\mathrm{Cosh}[z]^m}\right)^{1/m}}{m\left(1 - \left(\frac{\mathrm{Sinh}[z]^m}{\mathrm{Cosh}[z]^m}\right)^{2/m}\right)} \partial_z \frac{\mathrm{Sinh}[z]^m}{\mathrm{Cosh}[z]^m}$
- **Note: This rule should be replaced with a more general one.**
- **Rule: If $\frac{1}{m} \in \mathbb{Z} \bigwedge \frac{1}{m} > 1$, then**

$$\int \frac{\mathrm{Sinh}[a + b\,x]^m}{\mathrm{Cosh}[a + b\,x]^m} \, dx \longrightarrow \frac{1}{b\,m} \, \mathrm{Subst}\left[\int \frac{x^{1/m}}{1 - x^{2/m}} \, dx, \, x, \, \frac{\mathrm{Sinh}[a + b\,x]^m}{\mathrm{Cosh}[a + b\,x]^m}\right]$$

- **Program code:**

```
Int[Sinh[a_.+b_.*x_]^m_*Cosh[a_.+b_.*x_]^n_,x_Symbol] :=
  Dist[1/(b*m),Subst[Int[x^(1/m)/(1-x^(2/m)),x],x,Sinh[a+b*x]^m/Cosh[a+b*x]^m]] /;
FreeQ[{a,b},x] && ZeroQ[m+n] && IntegerQ[1/m] && 1/m>1
```

- **Basis: If $\frac{1}{n} \in \mathbb{Z}$, then** $\frac{\mathrm{Cosh}[z]^n}{\mathrm{Sinh}[z]^n} = \frac{\left(\frac{\mathrm{Cosh}[z]^n}{\mathrm{Sinh}[z]^n}\right)^{1/n}}{n\left(1 - \left(\frac{\mathrm{Cosh}[z]^n}{\mathrm{Sinh}[z]^n}\right)^{2/n}\right)} \partial_z \frac{\mathrm{Cosh}[z]^n}{\mathrm{Sinh}[z]^n}$

```
Int[Sinh[a_.+b_.*x_]^m_*Cosh[a_.+b_.*x_]^n_,x_Symbol] :=
  Dist[1/(b*n),Subst[Int[x^(1/n)/(1-x^(2/n)),x],x,Cosh[a+b*x]^n/Sinh[a+b*x]^n]] /;
FreeQ[{a,b},x] && ZeroQ[m+n] && IntegerQ[1/n] && 1/n>1
```

$$\int x^m \left(a + b \operatorname{Cosh}[d + e x]^2 + c \operatorname{Sinh}[d + e x]^2\right)^n dx$$

- **Derivation: Algebraic simplification**
- **Basis:** $a + b \operatorname{Cosh}[z]^2 + c \operatorname{Sinh}[z]^2 = \frac{1}{2} (2 a + b - c + (b + c) \operatorname{Cosh}[2 z])$
- **Rule: If** $m \in \mathbb{Z} \wedge m > 0 \wedge a + b \neq 0 \wedge a + c \neq 0$**, then**

$$\int \frac{x^m}{a + b \operatorname{Cosh}[d + e x]^2 + c \operatorname{Sinh}[d + e x]^2} dx \longrightarrow 2 \int \frac{x^m}{2 a + b - c + (b + c) \operatorname{Cosh}[2 d + 2 e x]} dx$$

- **Program code:**

```
Int[x_^m_./(a_.+b_.*Cosh[d_.+e_.*x_]^2+c_.*Sinh[d_.+e_.*x_]^2),x_Symbol] :=
  Dist[2,Int[x^m/(2*a+b-c+(b+c)*Cosh[2*d+2*e*x]),x]] /;
FreeQ[{a,b,c,d,e},x] && IntegerQ[m] && m>0 && NonzeroQ[a+b] && NonzeroQ[a+c]
```

$$\int x^m\,(a+b\,\text{Sinh}[c+d\,x]\,\text{Cosh}[c+d\,x])^n\,dx$$

- **Derivation: Algebraic simplification**
- **Basis:** $\text{Sinh}[z]\,\text{Cosh}[z]=\frac{1}{2}\,\text{Sinh}[2\,z]$
- **Rule: If** $m\in\mathbb{Z}\land m>0$, **then**

$$\int\frac{x^m}{a+b\,\text{Sinh}[c+d\,x]\,\text{Cosh}[c+d\,x]}\,dx\;\longrightarrow\;\int\frac{x^m}{a+\frac{1}{2}\,b\,\text{Sinh}[2\,c+2\,d\,x]}\,dx$$

- **Program code:**

```
Int[x_^m_./(a_+b_.*Sinh[c_.+d_.*x_]*Cosh[c_.+d_.*x_]),x_Symbol] :=
  Int[x^m/(a+b*Sinh[2*c+2*d*x]/2),x] /;
FreeQ[{a,b,c,d},x] && IntegerQ[m] && m>0
```

- **Derivation: Algebraic simplification**
- **Basis:** $\text{Sinh}[z]\,\text{Cosh}[z]=\frac{1}{2}\,\text{Sinh}[2\,z]$
- **Rule: If** $n-\frac{1}{2}\in\mathbb{Z}$, **then**

$$\int(a+b\,\text{Sinh}[c+d\,x]\,\text{Cosh}[c+d\,x])^n\,dx\;\longrightarrow\;\int\left(a+\frac{1}{2}\,b\,\text{Sinh}[2\,c+2\,d\,x]\right)^n\,dx$$

- **Program code:**

```
Int[(a_+b_.*Sinh[c_.+d_.*x_]*Cosh[c_.+d_.*x_])^n_,x_Symbol] :=
  Int[(a+b*Sinh[2*c+2*d*x]/2)^n,x] /;
FreeQ[{a,b,c,d},x] && HalfIntegerQ[n]
```

$$\int \mathrm{Sinh}[a + b\, x^n]^p\, \mathrm{Cosh}[a + b\, x^n]^p\, dx$$

- **Derivation: Algebraic simplification**

- **Basis:** $\mathrm{Sinh}[z]\, \mathrm{Cosh}[z] = \frac{1}{2}\, \mathrm{Sinh}[2\, z]$

- **Rule: If** $n,\ p \in \mathbb{Z}$**, then**

$$\int \mathrm{Sinh}[a + b\, x^n]^p\, \mathrm{Cosh}[a + b\, x^n]^p\, dx \longrightarrow \frac{1}{2} \int \mathrm{Sinh}[2\, a + 2\, b\, x^n]^p\, dx$$

- **Program code:**

```
Int[Sinh[a_.+b_.*x_^n_]^p_.*Cosh[a_.+b_.*x_^n_]^p_.,x_Symbol] :=
  Dist[1/2,Int[Sinh[2*a+2*b*x^n]^p,x]] /;
FreeQ[{a,b},x] && IntegersQ[n,p]
```

$$\int (a\,\text{Csch}[c+d\,x]+a\,\text{Sinh}[c+d\,x])^n\,dx$$

- **Derivation: Algebraic simplification**
- **Basis:** $\text{Csch}[z]+\text{Sinh}[z]=\text{Cosh}[z]\,\text{Coth}[z]$
- **Rule:**

$$\int (a\,\text{Csch}[c+d\,x]+a\,\text{Sinh}[c+d\,x])^n\,dx \longrightarrow \int (a\,\text{Cosh}[c+d\,x]\,\text{Coth}[c+d\,x])^n\,dx$$

- **Program code:**

```
Int[(a_.*Csch[c_.+d_.*x_]+b_.*Sinh[c_.+d_.*x_])^n_,x_Symbol] :=
  Int[(a*Cosh[c+d*x]*Coth[c+d*x])^n,x] /;
FreeQ[{a,b,c,d,n},x] && ZeroQ[a-b]
```

```
Int[(a_.*Sech[c_.+d_.*x_]+b_.*Cosh[c_.+d_.*x_])^n_,x_Symbol] :=
  Int[(-a*Sinh[c+d*x]*Tanh[c+d*x])^n,x] /;
FreeQ[{a,b,c,d,n},x] && ZeroQ[a+b]
```

$$\int \text{Sech}[v]^m\,(a+b\,\text{Tanh}[v])^n\,dx$$

- **Derivation: Algebraic simplification**
- **Basis:** $\frac{a+b\,\text{Tanh}[z]}{\text{Sech}[z]} = a\,\text{Cosh}[z] + b\,\text{Sinh}[z]$
- **Rule: If** $m,\ n \in \mathbb{Z} \land m+n=0 \bigwedge \frac{m-1}{2} \in \mathbb{Z}$**, then**

$$\int \text{Sech}[v]^m\,(a+b\,\text{Tanh}[v])^n\,dx \longrightarrow \int (a\,\text{Cosh}[v]+b\,\text{Sinh}[v])^n\,dx$$

- **Program code:**

```
Int[Sech[v_]^m_.*(a_+b_.*Tanh[v_])^n_., x_Symbol] :=
  Int[(a*Cosh[v]+b*Sinh[v])^n,x] /;
FreeQ[{a,b},x] && IntegersQ[m,n] && m+n==0 && OddQ[m]
```

```
Int[Csch[v_]^m_.*(a_+b_.*Coth[v_])^n_., x_Symbol] :=
  Int[(b*Cosh[v]+a*Sinh[v])^n,x] /;
FreeQ[{a,b},x] && IntegersQ[m,n] && m+n==0 && OddQ[m]
```

$$\int x^m \operatorname{Csch}[a+b x]^n \operatorname{Sech}[a+b x]^p \, dx$$

- **Derivation: Algebraic simplification**
- **Basis:** $\operatorname{Csch}[z] \operatorname{Sech}[z] = 2 \operatorname{Csch}[2 z]$
- **Rule: If** $n \in \mathbb{Z}$, **then**

$$\int x^m \operatorname{Csch}[a+b x]^n \operatorname{Sech}[a+b x]^n \, dx \longrightarrow 2^n \int x^m \operatorname{Csch}[2 a+2 b x]^n \, dx$$

- **Program code:**

```
Int[x_^m_.*Csch[a_.+b_.*x_]^n_.*Sech[a_.+b_.*x_]^n_., x_Symbol] :=
  Dist[2^n,Int[x^m*Csch[2*a+2*b*x]^n,x]] /;
FreeQ[{a,b},x] && RationalQ[m] && IntegerQ[n]
```

- **Derivation: Integration by parts**
- **Rule: If** $n, p \in \mathbb{Z} \land m > 0 \land n \neq p$, **then**

$$\int x^m \operatorname{Csch}[a+b x]^n \operatorname{Sech}[a+b x]^p \, dx \longrightarrow$$
$$x^m \int \operatorname{Csch}[a+b x]^n \operatorname{Sech}[a+b x]^p \, dx - m \int x^{m-1} \left(\int \operatorname{Csch}[a+b x]^n \operatorname{Sech}[a+b x]^p \, dx \right) dx$$

- **Program code:**

```
Int[x_^m_.*Csch[a_.+b_.*x_]^n_.*Sech[a_.+b_.*x_]^p_., x_Symbol] :=
  Module[{u=Block[{ShowSteps=False,StepCounter=Null}, Int[Csch[a+b*x]^n*Sech[a+b*x]^p,x]]},
  x^m*u - Dist[m,Int[x^(m-1)*u,x]]] /;
FreeQ[{a,b},x] && RationalQ[m] && IntegersQ[n,p] && m>0 && n≠p
```

$$\int u\,(a\,\text{Tanh}[v]^m+b\,\text{Sech}[v]^m)^n\,dx$$

- **Derivation: Algebraic simplification**
- **Basis: If** $a^2+b^2=0$, **then** $a\,\text{Tanh}[z]+b\,\text{Sech}[z]=a\,\text{Tanh}\left[\frac{z}{2}-\frac{\pi}{4}\,\frac{a}{b}\right]$
- **Rule: If** $a^2-b^2=0\ \bigwedge\ \frac{n}{2}\in\mathbb{Z}$, **then**

$$\int(a\,\text{Tanh}[v]+b\,\text{Sech}[v])^n\,dx\ \longrightarrow\ a^n\int\text{Tanh}\left[\frac{v}{2}-\frac{\pi}{4}\,\frac{a}{b}\right]^n\,dx$$

- **Program code:**

```
Int[(a_.*Tanh[v_]+b_.*Sech[v_])^n_,x_Symbol] :=
  Dist[a^n,Int[Tanh[v/2-Pi/4*a/b]^n,x]] /;
FreeQ[{a,b},x] && ZeroQ[a^2+b^2] && EvenQ[n]
```

- **Basis: If** $a^2-b^2=0$, **then** $a\,\text{Coth}[z]+b\,\text{Csch}[z]=a\,\text{Coth}\left[\frac{z}{2}+\frac{\pi i}{4}\,\frac{a-b}{b}\right]$

```
Int[(a_.*Coth[v_]+b_.*Csch[v_])^n_,x_Symbol] :=
  Dist[a^n,Int[Coth[v/2+Pi*I/4*(a-b)/b]^n,x]] /;
FreeQ[{a,b},x] && ZeroQ[a^2-b^2] && EvenQ[n]
```

- **Derivation: Algebraic simplification**
- **Basis:** $a\,\text{Sech}[z]+b\,\text{Tanh}[z]=\frac{a+b\,\text{Sinh}[z]}{\text{Cosh}[z]}$
- **Rule: If** $m,\,n\in\mathbb{Z}\ \bigwedge\ \left(\frac{m\,n-1}{2}\in\mathbb{Z}\ \bigvee\ m\,n<0\right)\ \bigwedge\ \neg\,(m=2\ \land\ a-b=0)$, **then**

$$\int u\,(a\,\text{Tanh}[v]^m+b\,\text{Sech}[v]^m)^n\,dx\ \longrightarrow\ \int\frac{u\,(a+b\,\text{Sinh}[v]^m)^n}{\text{Cosh}[v]^{m\,n}}\,dx$$

- **Program code:**

```
Int[u_.*(a_.*Sech[v_]^m_.+b_.*Tanh[v_]^m_.)^n_.,x_Symbol] :=
  Int[u*(a+b*Sinh[v]^m)^n/Cosh[v]^(m*n),x] /;
FreeQ[{a,b},x] && IntegersQ[m,n] && (OddQ[m*n] || m*n<0) && Not[m==2 && ZeroQ[a-b]]
```

```
Int[u_.*(a_.*Csch[v_]^m_.+b_.*Coth[v_]^m_.)^n_.,x_Symbol] :=
  Int[u*(a+b*Cosh[v]^m)^n/Sinh[v]^(m*n),x] /;
FreeQ[{a,b},x] && IntegersQ[m,n] && (OddQ[m*n] || m*n<0) && Not[m==2 && ZeroQ[a+b]]
```