$$\int \text{ArcTanh}[a\,x]^n\,dx$$

- **Reference: CRC 585, A&S 4.6.45**
- **Derivation: Integration by parts**
- **Rule:**

$$\int \text{ArcTanh}[a\,x]\,dx \longrightarrow x\,\text{ArcTanh}[a\,x] + \frac{\text{Log}\left[1-a^2\,x^2\right]}{2\,a}$$

- **Program code:**

```
Int[ArcTanh[a_.*x_],x_Symbol] :=
  x*ArcTanh[a*x] + Log[1-a^2*x^2]/(2*a) /;
FreeQ[a,x]
```

- **Derivation: Integration by parts**
- **Rule: If $n \in \mathbb{Z} \wedge n > 1$, then**

$$\int \text{ArcTanh}[a\,x]^n\,dx \longrightarrow x\,\text{ArcTanh}[a\,x]^n - a\,n\int \frac{x\,\text{ArcTanh}[a\,x]^{n-1}}{1-a^2\,x^2}\,dx$$

- **Program code:**

```
Int[ArcTanh[a_.*x_]^n_,x_Symbol] :=
  x*ArcTanh[a*x]^n -
  Dist[a*n,Int[x*ArcTanh[a*x]^(n-1)/(1-a^2*x^2),x]] /;
FreeQ[a,x] && IntegerQ[n] && n>1
```

$$\int x^m \operatorname{ArcTanh}[a\,x]^n \, dx$$

- **Derivation: Iterated integration by parts**
- **Rule: If $n \in \mathbb{Z} \wedge n > 0$, then**

$$\int x \operatorname{ArcTanh}[a\,x]^n \, dx \longrightarrow -\frac{\operatorname{ArcTanh}[a\,x]^n}{2\,a^2} + \frac{x^2 \operatorname{ArcTanh}[a\,x]^n}{2} + \frac{n}{2\,a} \int \operatorname{ArcTanh}[a\,x]^{n-1} \, dx$$

- **Program code:**

```
Int[x_*ArcTanh[a_.*x_]^n_.,x_Symbol] :=
  -ArcTanh[a*x]^n/(2*a^2) + x^2*ArcTanh[a*x]^n/2 +
  Dist[n/(2*a),Int[ArcTanh[a*x]^(n-1),x]] /;
FreeQ[a,x] && IntegerQ[n] && n>0
```

- **Derivation: Iterated integration by parts**
- **Rule: If $n \in \mathbb{Z} \wedge n > 0 \wedge m > 1$, then**

$$\int x^m \operatorname{ArcTanh}[a\,x]^n \, dx \longrightarrow -\frac{x^{m-1} \operatorname{ArcTanh}[a\,x]^n}{a^2\,(m+1)} + \frac{x^{m+1} \operatorname{ArcTanh}[a\,x]^n}{m+1} + \frac{n}{a\,(m+1)} \int x^{m-1} \operatorname{ArcTanh}[a\,x]^{n-1} \, dx + \frac{m-1}{a^2\,(m+1)} \int x^{m-2} \operatorname{ArcTanh}[a\,x]^n \, dx$$

- **Program code:**

```
Int[x_^m_*ArcTanh[a_.*x_]^n_.,x_Symbol] :=
  -x^(m-1)*ArcTanh[a*x]^n/(a^2*(m+1)) + x^(m+1)*ArcTanh[a*x]^n/(m+1) +
  Dist[n/(a*(m+1)),Int[x^(m-1)*ArcTanh[a*x]^(n-1),x]] +
  Dist[(m-1)/(a^2*(m+1)),Int[x^(m-2)*ArcTanh[a*x]^n,x]] /;
FreeQ[a,x] && IntegerQ[n] && n>0 && RationalQ[m] && m>1
```

- **Derivation: Integration by parts**
- **Rule: If $n \in \mathbb{Z} \wedge n > 1$, then**

$$\int \frac{\text{ArcTanh}[a\,x]^n}{x}\,dx \longrightarrow 2\,\text{ArcTanh}[a\,x]^n\,\text{ArcTanh}\left[1-\frac{2}{1-a\,x}\right]-2\,a\,n\int \frac{\text{ArcTanh}[a\,x]^{n-1}\,\text{ArcTanh}\left[1-\frac{2}{1-a\,x}\right]}{1-a^2\,x^2}\,dx$$

- **Program code:**

```
Int[ArcTanh[a_.*x_]^n_/x_,x_Symbol] :=
  2*ArcTanh[a*x]^n*ArcTanh[1-2/(1-a*x)] -
  Dist[2*a*n,Int[ArcTanh[a*x]^(n-1)*ArcTanh[1-2/(1-a*x)]/(1-a^2*x^2),x]] /;
FreeQ[a,x] && IntegerQ[n] && n>1
```

- **Derivation: Integration by parts**
- **Rule: If $n \in \mathbb{Z} \wedge n > 0$, then**

$$\int \frac{\text{ArcTanh}[a\,x]^n}{x^2}\,dx \longrightarrow -\frac{\text{ArcTanh}[a\,x]^n}{x}+a\,n\int \frac{\text{ArcTanh}[a\,x]^{n-1}}{x\,\left(1-a^2\,x^2\right)}\,dx$$

- **Program code:**

```
Int[ArcTanh[a_.*x_]^n_./x_^2,x_Symbol] :=
  -ArcTanh[a*x]^n/x +
  Dist[a*n,Int[ArcTanh[a*x]^(n-1)/(x*(1-a^2*x^2)),x]] /;
FreeQ[a,x] && IntegerQ[n] && n>0
```

- **Derivation: Inverted iterated integration by parts**
- **Rule: If $n \in \mathbb{Z} \wedge n > 0$, then**

$$\int \frac{\text{ArcTanh}[a\,x]^n}{x^3}\,dx \longrightarrow \frac{a^2\,\text{ArcTanh}[a\,x]^n}{2}-\frac{\text{ArcTanh}[a\,x]^n}{2\,x^2}+\frac{a\,n}{2}\int \frac{\text{ArcTanh}[a\,x]^{n-1}}{x^2}\,dx$$

- **Program code:**

```
Int[ArcTanh[a_.*x_]^n_./x_^3,x_Symbol] :=
  a^2*ArcTanh[a*x]^n/2 - ArcTanh[a*x]^n/(2*x^2) +
  Dist[a*n/2,Int[ArcTanh[a*x]^(n-1)/x^2,x]] /;
FreeQ[a,x] && IntegerQ[n] && n>0
```

- **Derivation: Inverted iterated integration by parts**

- **Rule: If $n \in \mathbb{Z} \bigwedge n > 0 \bigwedge m < -3$, then**

$$\int x^m \operatorname{ArcTanh}[a\,x]^n\,dx \longrightarrow \frac{x^{m+1}\operatorname{ArcTanh}[a\,x]^n}{m+1} - \frac{a^2\,x^{m+3}\operatorname{ArcTanh}[a\,x]^n}{m+1} - \frac{a\,n}{m+1}\int x^{m+1}\operatorname{ArcTanh}[a\,x]^{n-1}\,dx + \frac{a^2\,(m+3)}{m+1}\int x^{m+2}\operatorname{ArcTanh}[a\,x]^n\,dx$$

- **Program code:**

```
Int[x_^m_*ArcTanh[a_.*x_]^n_.,x_Symbol] :=
  x^(m+1)*ArcTanh[a*x]^n/(m+1) - a^2*x^(m+3)*ArcTanh[a*x]^n/(m+1) -
  Dist[a*n/(m+1),Int[x^(m+1)*ArcTanh[a*x]^(n-1),x]] +
  Dist[a^2*(m+3)/(m+1),Int[x^(m+2)*ArcTanh[a*x]^n,x]] /;
FreeQ[a,x] && IntegerQ[n] && n>0 && RationalQ[m] && m<-3
```

$$\int \frac{\text{ArcTanh}[a\,x]^n}{c+d\,x}\,dx$$

- **Derivation: Integration by parts**

- **Rule: If** $a^2\,c^2=d^2\,\wedge\,n\in\mathbb{Z}\,\wedge\,n>0$, **then**

$$\int \frac{\text{ArcTanh}[a\,x]^n}{c+d\,x}\,dx\,\longrightarrow\,-\frac{\text{ArcTanh}[a\,x]^n\,\text{Log}\left[\frac{2\,c}{c+d\,x}\right]}{d}+\frac{a\,n}{d}\int\frac{\text{ArcTanh}[a\,x]^{n-1}\,\text{Log}\left[\frac{2\,c}{c+d\,x}\right]}{1-a^2\,x^2}\,dx$$

- **Program code:**

```
Int[ArcTanh[a_.*x_]^n_./(c_+d_.*x_),x_Symbol] :=
  -ArcTanh[a*x]^n*Log[2*c/(c+d*x)]/d +
  Dist[a*n/d,Int[ArcTanh[a*x]^(n-1)*Log[2*c/(c+d*x)]/(1-a^2*x^2),x]] /;
FreeQ[{a,c,d},x] && ZeroQ[a^2*c^2-d^2] && IntegerQ[n] && n>0
```

$$\int \frac{x^m \operatorname{ArcTanh}[a\,x]^n}{c+d\,x}\,dx$$

- **Derivation: Integration by parts**

- **Rule: If** $a^2 c^2 = d^2 \wedge n \in \mathbb{Z} \wedge n > 0$**, then**

$$\int \frac{\operatorname{ArcTanh}[a\,x]^n}{x\,(c+d\,x)}\,dx \longrightarrow \frac{\operatorname{ArcTanh}[a\,x]^n \operatorname{Log}\left[2-\frac{2\,c}{c+d\,x}\right]}{c} - \frac{a\,n}{c}\int \frac{\operatorname{ArcTanh}[a\,x]^{n-1} \operatorname{Log}\left[2-\frac{2\,c}{c+d\,x}\right]}{1-a^2\,x^2}\,dx$$

- **Program code:**

```
Int[ArcTanh[a_.*x_]^n_./(x_*(c_+d_.*x_)),x_Symbol] :=
  ArcTanh[a*x]^n*Log[2-2*c/(c+d*x)]/c -
  Dist[a*n/c,Int[ArcTanh[a*x]^(n-1)*Log[2-2*c/(c+d*x)]/(1-a^2*x^2),x]] /;
FreeQ[{a,c,d},x] && ZeroQ[a^2*c^2-d^2] && IntegerQ[n] && n>0
```

- **Derivation: Integration by parts**

- **Rule: If** $a^2 c^2 = d^2 \wedge n \in \mathbb{Z} \wedge n > 0$**, then**

$$\int \frac{\operatorname{ArcTanh}[a\,x]^n}{c\,x+d\,x^2}\,dx \longrightarrow \frac{\operatorname{ArcTanh}[a\,x]^n \operatorname{Log}\left[2-\frac{2\,c}{c+d\,x}\right]}{c} - \frac{a\,n}{c}\int \frac{\operatorname{ArcTanh}[a\,x]^{n-1} \operatorname{Log}\left[2-\frac{2\,c}{c+d\,x}\right]}{1-a^2\,x^2}\,dx$$

- **Program code:**

```
Int[ArcTanh[a_.*x_]^n_./(c_.*x_+d_.*x_^2),x_Symbol] :=
  ArcTanh[a*x]^n*Log[2-2*c/(c+d*x)]/c -
  Dist[a*n/c,Int[ArcTanh[a*x]^(n-1)*Log[2-2*c/(c+d*x)]/(1-a^2*x^2),x]] /;
FreeQ[{a,c,d},x] && ZeroQ[a^2*c^2-d^2] && IntegerQ[n] && n>0
```

- **Derivation: Algebraic simplification**
- **Basis:** $\frac{x}{c+d\,x} == \frac{1}{d} - \frac{c}{d\,(c+d\,x)}$
- **Rule: If** $a^2\,c^2 = d^2 \wedge m > 0 \wedge n \in \mathbb{Z} \wedge n > 0$**, then**

$$\int \frac{x^m\,\text{ArcTanh}[a\,x]^n}{c+d\,x}\,dx \longrightarrow \frac{1}{d}\int x^{m-1}\,\text{ArcTanh}[a\,x]^n\,dx - \frac{c}{d}\int \frac{x^{m-1}\,\text{ArcTanh}[a\,x]^n}{c+d\,x}\,dx$$

- **Program code:**

```
Int[x_^m_.*ArcTanh[a_.*x_]^n_./(c_+d_.*x_),x_Symbol] :=
  Dist[1/d,Int[x^(m-1)*ArcTanh[a*x]^n,x]] -
  Dist[c/d,Int[x^(m-1)*ArcTanh[a*x]^n/(c+d*x),x]] /;
FreeQ[{a,c,d},x] && ZeroQ[a^2*c^2-d^2] && RationalQ[m] && m>0 && IntegerQ[n] && n>0
```

- **Derivation: Algebraic simplification**
- **Basis:** $\frac{1}{c+d\,x} == \frac{1}{c} - \frac{d\,x}{c\,(c+d\,x)}$
- **Rule: If** $a^2\,c^2 = d^2 \wedge m < -1 \wedge n \in \mathbb{Z} \wedge n > 0$**, then**

$$\int \frac{x^m\,\text{ArcTanh}[a\,x]^n}{c+d\,x}\,dx \longrightarrow \frac{1}{c}\int x^m\,\text{ArcTanh}[a\,x]^n\,dx - \frac{d}{c}\int \frac{x^{m+1}\,\text{ArcTanh}[a\,x]^n}{c+d\,x}\,dx$$

- **Program code:**

```
Int[x_^m_*ArcTanh[a_.*x_]^n_./(c_+d_.*x_),x_Symbol] :=
  Dist[1/c,Int[x^m*ArcTanh[a*x]^n,x]] -
  Dist[d/c,Int[x^(m+1)*ArcTanh[a*x]^n/(c+d*x),x]] /;
FreeQ[{a,c,d},x] && ZeroQ[a^2*c^2-d^2] && RationalQ[m] && m<-1 && IntegerQ[n] && n>0
```

$$\int \frac{\text{ArcTanh}[a\,x]^n}{c+d\,x^2}\,dx$$

- **Derivation: Reciprocal rule for integration**

- **Rule: If $a^2 c + d = 0$, then**

$$\int \frac{1}{(c+d\,x^2)\,\text{ArcTanh}[a\,x]}\,dx \longrightarrow \frac{\text{Log}[\text{ArcTanh}[a\,x]]}{a\,c}$$

- **Program code:**

```
Int[1/((c_+d_.*x_^2)*ArcTanh[a_.*x_]),x_Symbol] :=
  Log[ArcTanh[a*x]]/(a*c) /;
FreeQ[{a,c,d},x] && ZeroQ[a^2*c+d]
```

- **Derivation: Power rule for integration**

- **Rule: If $a^2 c + d = 0 \wedge n \neq -1$, then**

$$\int \frac{\text{ArcTanh}[a\,x]^n}{c+d\,x^2}\,dx \longrightarrow \frac{\text{ArcTanh}[a\,x]^{n+1}}{a\,c\,(n+1)}$$

- **Program code:**

```
Int[ArcTanh[a_.*x_]^n_./(c_+d_.*x_^2),x_Symbol] :=
  ArcTanh[a*x]^(n+1)/(a*c*(n+1)) /;
FreeQ[{a,c,d,n},x] && ZeroQ[a^2*c+d] && NonzeroQ[n+1]
```

$$\int \frac{x^m \operatorname{ArcTanh}[a x]^n}{c + d x^2} \, dx$$

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{x}{1-a^2 x^2} = -\frac{1}{a\left(1-a^2 x^2\right)} + \frac{1}{a(1-a x)}$
- **Rule: If** $a^2 c + d = 0 \wedge n > 0$**, then**

$$\int \frac{x \operatorname{ArcTanh}[a x]^n}{c + d x^2} \, dx \longrightarrow \frac{\operatorname{ArcTanh}[a x]^{n+1}}{d(n+1)} + \frac{1}{a c} \int \frac{\operatorname{ArcTanh}[a x]^n}{1 - a x} \, dx$$

- **Program code:**

```
Int[x_*ArcTanh[a_.*x_]^n_./(c_+d_.*x_^2),x_Symbol] :=
  ArcTanh[a*x]^(n+1)/(d*(n+1)) +
  Dist[1/(a*c),Int[ArcTanh[a*x]^n/(1-a*x),x]] /;
FreeQ[{a,c,d},x] && ZeroQ[a^2*c+d] && RationalQ[n] && n>0
```

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{1}{x\left(1-a^2 x^2\right)} = \frac{a}{1-a^2 x^2} + \frac{1}{x(1+a x)}$
- **Rule: If** $a^2 c + d = 0 \wedge n > 0$**, then**

$$\int \frac{\operatorname{ArcTanh}[a x]^n}{x\left(c + d x^2\right)} \, dx \longrightarrow \frac{\operatorname{ArcTanh}[a x]^{n+1}}{c(n+1)} + \frac{1}{c} \int \frac{\operatorname{ArcTanh}[a x]^n}{x(1 + a x)} \, dx$$

- **Program code:**

```
Int[ArcTanh[a_.*x_]^n_./(x_*(c_+d_.*x_^2)),x_Symbol] :=
  ArcTanh[a*x]^(n+1)/(c*(n+1)) +
  Dist[1/c,Int[ArcTanh[a*x]^n/(x*(1+a*x)),x]] /;
FreeQ[{a,c,d},x] && ZeroQ[a^2*c+d] && RationalQ[n] && n>0
```

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{1}{x\left(1-a^2 x^2\right)} = \frac{a}{1-a^2 x^2} + \frac{1}{x\,(1+a\,x)}$
- **Rule: If** $a^2 c + d = 0 \land n > 0$**, then**

$$\int \frac{\text{ArcTanh}[a\,x]^n}{c\,x + d\,x^3}\,dx \longrightarrow \frac{\text{ArcTanh}[a\,x]^{n+1}}{c\,(n+1)} + \frac{1}{c}\int \frac{\text{ArcTanh}[a\,x]^n}{x\,(1+a\,x)}\,dx$$

- **Program code:**

```
Int[ArcTanh[a_.*x_]^n_./(c_.*x_+d_.*x_^3),x_Symbol] :=
  ArcTanh[a*x]^(n+1)/(c*(n+1)) +
  Dist[1/c,Int[ArcTanh[a*x]^n/(x*(1+a*x)),x]] /;
FreeQ[{a,c,d},x] && ZeroQ[a^2*c+d] && RationalQ[n] && n>0
```

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{x^2}{c+d\,x^2} = \frac{1}{d} - \frac{c}{d\left(c+d\,x^2\right)}$
- **Rule: If** $a^2 c + d = 0 \land m > 1 \land n > 0$ **, then**

$$\int \frac{x^m\,\text{ArcTanh}[a\,x]^n}{c + d\,x^2}\,dx \longrightarrow \frac{1}{d}\int x^{m-2}\,\text{ArcTanh}[a\,x]^n\,dx - \frac{c}{d}\int \frac{x^{m-2}\,\text{ArcTanh}[a\,x]^n}{c + d\,x^2}\,dx$$

- **Program code:**

```
Int[x_^m_*ArcTanh[a_.*x_]^n_./(c_+d_.*x_^2),x_Symbol] :=
  Dist[1/d,Int[x^(m-2)*ArcTanh[a*x]^n,x]] -
  Dist[c/d,Int[x^(m-2)*ArcTanh[a*x]^n/(c+d*x^2),x]] /;
FreeQ[{a,c,d},x] && ZeroQ[a^2*c+d] && RationalQ[{m,n}] && m>1 && n>0
```

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{1}{c+d\,x^2} = \frac{1}{c} - \frac{d\,x^2}{c\left(c+d\,x^2\right)}$
- **Rule: If** $a^2 c + d = 0 \land m < -1 \land n > 0$**, then**

$$\int \frac{x^m\,\text{ArcTanh}[a\,x]^n}{c + d\,x^2}\,dx \longrightarrow \frac{1}{c}\int x^m\,\text{ArcTanh}[a\,x]^n\,dx - \frac{d}{c}\int \frac{x^{m+2}\,\text{ArcTanh}[a\,x]^n}{c + d\,x^2}\,dx$$

- **Program code:**

```
Int[x_^m_*ArcTanh[a_.*x_]^n_./(c_+d_.*x_^2),x_Symbol] :=
  Dist[1/c,Int[x^m*ArcTanh[a*x]^n,x]] -
  Dist[d/c,Int[x^(m+2)*ArcTanh[a*x]^n/(c+d*x^2),x]] /;
FreeQ[{a,c,d},x] && ZeroQ[a^2*c+d] && RationalQ[{m,n}] && m<-1 && n>0
```

- **Derivation: Integration by substitution**
- **Basis: If** $m \in \mathbb{Z}$ **or** $a > 0$, $\frac{x^m \operatorname{ArcTanh}[a x]^n}{1-a^2 x^2} = \frac{\operatorname{Tanh}[\operatorname{ArcTanh}[a x]]^m \operatorname{ArcTanh}[a x]^n}{a^{m+1}} \partial_x \operatorname{ArcTanh}[a x]$
- **Rule: If** $a^2 c + d = 0 \wedge m, n \in \mathbb{Q} \wedge (n < 0 \vee n \notin \mathbb{Z}) \wedge (m \in \mathbb{Z} \vee a > 0)$, **then**

$$\int \frac{x^m \operatorname{ArcTanh}[a x]^n}{c + d x^2} \, dx \longrightarrow \frac{1}{a^{m+1} c} \operatorname{Subst}\left[\int x^n \operatorname{Tanh}[x]^m \, dx, x, \operatorname{ArcTanh}[a x]\right]$$

- **Program code:**

```
Int[x_^m_.*ArcTanh[a_.*x_]^n_/(c_+d_.*x_^2),x_Symbol] :=
  Dist[1/(a^(m+1)*c),Subst[Int[x^n*Tanh[x]^m,x],x,ArcTanh[a*x]]] /;
FreeQ[{a,c,d},x] && ZeroQ[a^2*c+d] && RationalQ[{m,n}] && (n<0 || Not[IntegerQ[n]]) && (IntegerQ[m]
```

- **Derivation: Integration by substitution**
- **Basis:** $\frac{x^m \operatorname{ArcTanh}[a x]^n}{1-a^2 x^2} = \frac{1}{a} \left(\frac{\operatorname{Tanh}[\operatorname{ArcTanh}[a x]]}{a}\right)^m \operatorname{ArcTanh}[a x]^n \partial_x \operatorname{ArcTanh}[a x]$
- **Rule: If** $a^2 c + d = 0 \wedge m, n \in \mathbb{Q} \wedge (n < 0 \vee n \notin \mathbb{Z}) \wedge \neg (m \in \mathbb{Z} \vee a > 0)$, **then**

$$\int \frac{x^m \operatorname{ArcTanh}[a x]^n}{c + d x^2} \, dx \longrightarrow \frac{1}{a c} \operatorname{Subst}\left[\int x^n \left(\frac{\operatorname{Tanh}[x]}{a}\right)^m dx, x, \operatorname{ArcTanh}[a x]\right]$$

- **Program code:**

```
Int[x_^m_.*ArcTanh[a_.*x_]^n_/(c_+d_.*x_^2),x_Symbol] :=
  Dist[1/(a*c),Subst[Int[x^n*(Tanh[x]/a)^m,x],x,ArcTanh[a*x]]] /;
FreeQ[{a,c,d},x] && ZeroQ[a^2*c+d] && RationalQ[{m,n}] && (n<0 || Not[IntegerQ[n]]) && Not[IntegerQ[
```

$$\int \frac{\text{ArcTanh}[a\,x]^n\,\text{ArcTanh}[u]}{c+d\,x^2}\,dx$$

- **Derivation: Algebraic simplification**
- **Basis:** $\text{ArcTanh}[z] = \frac{1}{2}\,\text{Log}[1+z] - \frac{1}{2}\,\text{Log}[1-z]$
- **Rule: If** $a^2\,c+d=0\,\bigwedge\,n>0\,\bigwedge\,\left(u^2=\left(1-\frac{2}{1+a\,x}\right)^2\,\bigvee\,u^2=\left(1-\frac{2}{1-a\,x}\right)^2\right)$, **then**

$$\int \frac{\text{ArcTanh}[a\,x]^n\,\text{ArcTanh}[u]}{c+d\,x^2}\,dx \;\longrightarrow\; \frac{1}{2}\int \frac{\text{ArcTanh}[a\,x]^n\,\text{Log}[1+u]}{c+d\,x^2}\,dx - \frac{1}{2}\int \frac{\text{ArcTanh}[a\,x]^n\,\text{Log}[1-u]}{c+d\,x^2}\,dx$$

- **Program code:**

```
Int[ArcTanh[a_.*x_]^n_.*ArcTanh[u_]/(c_+d_.*x_^2),x_Symbol] :=
  Dist[1/2,Int[ArcTanh[a*x]^n*Log[1+u]/(c+d*x^2),x]] -
  Dist[1/2,Int[ArcTanh[a*x]^n*Log[1-u]/(c+d*x^2),x]] /;
FreeQ[{a,c,d},x] && ZeroQ[a^2*c+d] && RationalQ[n] && n>0 && (ZeroQ[u^2-(1-2/(1+a*x))^2] || ZeroQ[u^2
```

$$\int \frac{\text{ArcTanh}[a\,x]^n\,\text{Log}[u]}{c+d\,x^2}\,dx$$

- **Derivation: Integration by parts**
- **Rule: If** $a^2\,c+d=0\,\wedge\,n>0\,\wedge\,(1-u)^2=\left(1-\frac{2}{1+a\,x}\right)^2$**, then**

$$\int \frac{\text{ArcTanh}[a\,x]^n\,\text{Log}[u]}{c+d\,x^2}\,dx \longrightarrow \frac{\text{ArcTanh}[a\,x]^n\,\text{PolyLog}[2,\,1-u]}{2\,a\,c}-\frac{n}{2}\int \frac{\text{ArcTanh}[a\,x]^{n-1}\,\text{PolyLog}[2,\,1-u]}{c+d\,x^2}\,dx$$

- **Program code:**

```
Int[ArcTanh[a_.*x_]^n_.*Log[u_]/(c_+d_.*x_^2),x_Symbol] :=
  ArcTanh[a*x]^n*PolyLog[2,1-u]/(2*a*c) -
  Dist[n/2,Int[ArcTanh[a*x]^(n-1)*PolyLog[2,1-u]/(c+d*x^2),x]] /;
FreeQ[{a,c,d},x] && ZeroQ[a^2*c+d] && RationalQ[n] && n>0 && ZeroQ[(1-u)^2-(1-2/(1+a*x))^2]
```

- **Derivation: Integration by parts**
- **Rule: If** $a^2\,c+d=0\,\wedge\,n>0\,\wedge\,(1-u)^2=\left(1-\frac{2}{1-a\,x}\right)^2$**, then**

$$\int \frac{\text{ArcTanh}[a\,x]^n\,\text{Log}[u]}{c+d\,x^2}\,dx \longrightarrow -\frac{\text{ArcTanh}[a\,x]^n\,\text{PolyLog}[2,\,1-u]}{2\,a\,c}+\frac{n}{2}\int \frac{\text{ArcTanh}[a\,x]^{n-1}\,\text{PolyLog}[2,\,1-u]}{c+d\,x^2}\,dx$$

- **Program code:**

```
Int[ArcTanh[a_.*x_]^n_.*Log[u_]/(c_+d_.*x_^2),x_Symbol] :=
  -ArcTanh[a*x]^n*PolyLog[2,1-u]/(2*a*c) +
  Dist[n/2,Int[ArcTanh[a*x]^(n-1)*PolyLog[2,1-u]/(c+d*x^2),x]] /;
FreeQ[{a,c,d},x] && ZeroQ[a^2*c+d] && RationalQ[n] && n>0 && ZeroQ[(1-u)^2-(1-2/(1-a*x))^2]
```

$$\int \frac{\text{ArcTanh}[a\,x]^n\,\text{PolyLog}[p,\,u]}{c+d\,x^2}\,dx$$

- **Derivation: Integration by parts**
- **Rule: If** $a^2 c + d = 0\ \wedge\ n > 0\ \wedge\ u^2 = \left(1 - \frac{2}{1+a\,x}\right)^2$**, then**

$$\int \frac{\text{ArcTanh}[a\,x]^n\,\text{PolyLog}[p,\,u]}{c+d\,x^2}\,dx \longrightarrow -\frac{\text{ArcTanh}[a\,x]^n\,\text{PolyLog}[p+1,\,u]}{2\,a\,c} + \frac{n}{2}\int \frac{\text{ArcTanh}[a\,x]^{n-1}\,\text{PolyLog}[p+1,\,u]}{c+d\,x^2}\,dx$$

- **Program code:**

```
Int[ArcTanh[a_.*x_]^n_.*PolyLog[p_,u_]/(c_+d_.*x_^2),x_Symbol] :=
  -ArcTanh[a*x]^n*PolyLog[p+1,u]/(2*a*c) +
  Dist[n/2,Int[ArcTanh[a*x]^(n-1)*PolyLog[p+1,u]/(c+d*x^2),x]] /;
FreeQ[{a,c,d,p},x] && ZeroQ[a^2*c+d] && RationalQ[n] && n>0 && ZeroQ[u^2-(1-2/(1+a*x))^2]
```

- **Derivation: Integration by parts**
- **Rule: If** $a^2 c + d = 0\ \wedge\ n > 0\ \wedge\ u^2 = \left(1 - \frac{2}{1-a\,x}\right)^2$**, then**

$$\int \frac{\text{ArcTanh}[a\,x]^n\,\text{PolyLog}[p,\,u]}{c+d\,x^2}\,dx \longrightarrow \frac{\text{ArcTanh}[a\,x]^n\,\text{PolyLog}[p+1,\,u]}{2\,a\,c} - \frac{n}{2}\int \frac{\text{ArcTanh}[a\,x]^{n-1}\,\text{PolyLog}[p+1,\,u]}{c+d\,x^2}\,dx$$

- **Program code:**

```
Int[ArcTanh[a_.*x_]^n_.*PolyLog[p_,u_]/(c_+d_.*x_^2),x_Symbol] :=
  ArcTanh[a*x]^n*PolyLog[p+1,u]/(2*a*c) -
  Dist[n/2,Int[ArcTanh[a*x]^(n-1)*PolyLog[p+1,u]/(c+d*x^2),x]] /;
FreeQ[{a,c,d,p},x] && ZeroQ[a^2*c+d] && RationalQ[n] && n>0 && ZeroQ[u^2-(1-2/(1-a*x))^2]
```

$$\int \frac{\text{ArcCoth}[a\,x]^m\,\text{ArcTanh}[a\,x]^n}{c+d\,x^2}\,dx$$

- **Rule: If** $a^2 c + d = 0$**, then**

$$\int \frac{1}{\left(c+d\,x^2\right)\text{ArcCoth}[a\,x]\,\text{ArcTanh}[a\,x]}\,dx \longrightarrow \frac{-\text{Log}[\text{ArcCoth}[a\,x]]+\text{Log}[\text{ArcTanh}[a\,x]]}{a\,c\,\text{ArcCoth}[a\,x]-a\,c\,\text{ArcTanh}[a\,x]}$$

- **Program code:**

```
Int[1/((ArcCoth[a_.*x_]*ArcTanh[a_.*x_]*(c_+d_.*x_^2)),x_Symbol] :=
  (-Log[ArcCoth[a*x]]+Log[ArcTanh[a*x]])/(a*c*ArcCoth[a*x]-a*c*ArcTanh[a*x]) /;
FreeQ[{a,c,d},x] && ZeroQ[a^2*c+d]
```

- **Derivation: Integration by parts**
- **Rule: If** $a^2 c + d = 0 \land m, n \in \mathbb{Z} \land 0 < n \leq m$**, then**

$$\int \frac{\text{ArcCoth}[a\,x]^m\,\text{ArcTanh}[a\,x]^n}{c+d\,x^2}\,dx \longrightarrow$$
$$\frac{\text{ArcCoth}[a\,x]^{m+1}\,\text{ArcTanh}[a\,x]^n}{a\,c\,(m+1)} - \frac{n}{m+1}\int \frac{\text{ArcCoth}[a\,x]^{m+1}\,\text{ArcTanh}[a\,x]^{n-1}}{c+d\,x^2}\,dx$$

- **Program code:**

```
Int[ArcCoth[a_.*x_]^m_.*ArcTanh[a_.*x_]^n_./(c_+d_.*x_^2),x_Symbol] :=
  ArcCoth[a*x]^(m+1)*ArcTanh[a*x]^n/(a*c*(m+1)) -
  Dist[n/(m+1),Int[ArcCoth[a*x]^(m+1)*ArcTanh[a*x]^(n-1)/(c+d*x^2),x]] /;
FreeQ[{a,c,d},x] && ZeroQ[a^2*c+d] && IntegersQ[m,n] && 0<n≤m
```

$$\int \left(c+d\,x^2\right)^m \text{ArcTanh}[a\,x]^n\,dx$$

- **Rule: If** $a^2 c+d=0 \land c>0$, **then**

$$\int \frac{\text{ArcTanh}[a\,x]}{\sqrt{c+d\,x^2}}\,dx \longrightarrow -\frac{2\,\text{ArcTanh}[a\,x]\,\text{ArcTan}\left[\frac{\sqrt{1-a\,x}}{\sqrt{1+a\,x}}\right]}{a\sqrt{c}}-\frac{i\,\text{PolyLog}\left[2,-\frac{i\sqrt{1-a\,x}}{\sqrt{1+a\,x}}\right]}{a\sqrt{c}}+\frac{i\,\text{PolyLog}\left[2,\frac{i\sqrt{1-a\,x}}{\sqrt{1+a\,x}}\right]}{a\sqrt{c}}$$

- **Program code:**

```
Int[ArcTanh[a_.*x_]/Sqrt[c_+d_.*x_^2],x_Symbol] :=
  -2*ArcTanh[a*x]*ArcTan[Sqrt[1-a*x]/Sqrt[1+a*x]]/(a*Sqrt[c]) -
  I*PolyLog[2,-I*Sqrt[1-a*x]/Sqrt[1+a*x]]/(a*Sqrt[c]) +
  I*PolyLog[2,I*Sqrt[1-a*x]/Sqrt[1+a*x]]/(a*Sqrt[c]) /;
FreeQ[{a,c,d},x] && ZeroQ[a^2*c+d] && PositiveQ[c]
```

- **Basis:** $\partial_x \frac{\sqrt{1-a^2 x^2}}{\sqrt{c-c\,a^2 x^2}}=0$
- **Rule: If** $a^2 c+d=0 \land \neg(c>0)$, **then**

$$\int \frac{\text{ArcTanh}[a\,x]}{\sqrt{c+d\,x^2}}\,dx \longrightarrow \frac{\sqrt{1-a^2 x^2}}{\sqrt{c+d\,x^2}}\int \frac{\text{ArcTanh}[a\,x]}{\sqrt{1-a^2 x^2}}\,dx$$

- **Program code:**

```
Int[ArcTanh[a_.*x_]/Sqrt[c_+d_.*x_^2],x_Symbol] :=
  Sqrt[1-a^2*x^2]/Sqrt[c+d*x^2]*Int[ArcTanh[a*x]/Sqrt[1-a^2*x^2],x] /;
FreeQ[{a,c,d},x] && ZeroQ[a^2*c+d] && Not[PositiveQ[c]]
```

- **Rule: If** $a^2 c+d=0$, **then**

$$\int \frac{\text{ArcTanh}[a\,x]}{\left(c+d\,x^2\right)^{3/2}}\,dx \longrightarrow -\frac{1}{a\,c\sqrt{c+d\,x^2}}+\frac{x\,\text{ArcTanh}[a\,x]}{c\sqrt{c+d\,x^2}}$$

- **Program code:**

```
Int[ArcTanh[a_.*x_]/(c_+d_.*x_^2)^(3/2),x_Symbol] :=
  -1/(a*c*Sqrt[c+d*x^2]) +
  x*ArcTanh[a*x]/(c*Sqrt[c+d*x^2]) /;
FreeQ[{a,c,d},x] && ZeroQ[a^2*c+d]
```

- **Rule: If** $a^2 c + d = 0 \wedge n > 1$, **then**

$$\int \frac{\text{ArcTanh}[a x]^n}{\left(c + d x^2\right)^{3/2}} \, dx \longrightarrow -\frac{n \, \text{ArcTanh}[a x]^{n-1}}{a c \sqrt{c + d x^2}} + \frac{x \, \text{ArcTanh}[a x]^n}{c \sqrt{c + d x^2}} + n (n-1) \int \frac{\text{ArcTanh}[a x]^{n-2}}{\left(c + d x^2\right)^{3/2}} \, dx$$

- **Program code:**

```
Int[ArcTanh[a_.*x_]^n_/(c_+d_.*x_^2)^(3/2),x_Symbol] :=
  -n*ArcTanh[a*x]^(n-1)/(a*c*Sqrt[c+d*x^2]) +
  x*ArcTanh[a*x]^n/(c*Sqrt[c+d*x^2]) +
  Dist[n*(n-1),Int[ArcTanh[a*x]^(n-2)/(c+d*x^2)^(3/2),x]] /;
FreeQ[{a,c,d},x] && ZeroQ[a^2*c+d] && RationalQ[n] && n>1
```

- **Rule: If** $a^2 c + d = 0 \wedge n < -1 \wedge n \neq -2$, **then**

$$\int \frac{\text{ArcTanh}[a x]^n}{\left(c + d x^2\right)^{3/2}} \, dx \longrightarrow$$

$$\frac{\text{ArcTanh}[a x]^{n+1}}{a c (n+1) \sqrt{c + d x^2}} - \frac{x \, \text{ArcTanh}[a x]^{n+2}}{c (n+1) (n+2) \sqrt{c + d x^2}} + \frac{1}{(n+1) (n+2)} \int \frac{\text{ArcTanh}[a x]^{n+2}}{\left(c + d x^2\right)^{3/2}} \, dx$$

- **Program code:**

```
Int[ArcTanh[a_.*x_]^n_/(c_+d_.*x_^2)^(3/2),x_Symbol] :=
  ArcTanh[a*x]^(n+1)/(a*c*(n+1)*Sqrt[c+d*x^2]) -
  x*ArcTanh[a*x]^(n+2)/(c*(n+1)*(n+2)*Sqrt[c+d*x^2]) +
  Dist[1/((n+1)*(n+2)),Int[ArcTanh[a*x]^(n+2)/(c+d*x^2)^(3/2),x]] /;
FreeQ[{a,c,d},x] && ZeroQ[a^2*c+d] && RationalQ[n] && n<-1 && n≠-2
```

- **Rule: If** $a^2 c + d = 0 \wedge m > 0$, **then**

$$\int \left(c + d x^2\right)^m \text{ArcTanh}[a x] \, dx \longrightarrow$$

$$\frac{\left(c + d x^2\right)^m}{2 a m (2 m + 1)} + \frac{x \left(c + d x^2\right)^m \text{ArcTanh}[a x]}{(2 m + 1)} + \frac{2 c m}{2 m + 1} \int \left(c + d x^2\right)^{m-1} \text{ArcTanh}[a x] \, dx$$

- **Program code:**

```
Int[(c_+d_.*x_^2)^m_.*ArcTanh[a_.*x_],x_Symbol] :=
  (c+d*x^2)^m/(2*a*m*(2*m+1)) +
  x*(c+d*x^2)^m*ArcTanh[a*x]/(2*m+1) +
  Dist[2*c*m/(2*m+1),Int[(c+d*x^2)^(m-1)*ArcTanh[a*x],x]] /;
FreeQ[{a,c,d},x] && ZeroQ[a^2*c+d] && RationalQ[m] && m>0
```

- **Rule: If** $a^2 c + d = 0 \bigwedge m < -1 \bigwedge m \neq -\frac{3}{2}$, **then**

$$\int \left(c + d x^2\right)^m \text{ArcTanh}[a x] \, dx \longrightarrow$$

$$-\frac{\left(c + d x^2\right)^{m+1}}{4 a c (m+1)^2} - \frac{x \left(c + d x^2\right)^{m+1} \text{ArcTanh}[a x]}{2 c (m+1)} + \frac{2 m + 3}{2 c (m+1)} \int \left(c + d x^2\right)^{m+1} \text{ArcTanh}[a x] \, dx$$

- **Program code:**

```
Int[(c_+d_.*x_^2)^m_*ArcTanh[a_.*x_],x_Symbol] :=
  -(c+d*x^2)^(m+1)/(4*a*c*(m+1)^2) -
  x*(c+d*x^2)^(m+1)*ArcTanh[a*x]/(2*c*(m+1)) +
  Dist[(2*m+3)/(2*c*(m+1)),Int[(c+d*x^2)^(m+1)*ArcTanh[a*x],x]] /;
FreeQ[{a,c,d},x] && ZeroQ[a^2*c+d] && RationalQ[m] && m<-1 && m≠-3/2
```

- **Rule: If** $a^2 c + d = 0 \bigwedge m < -1 \bigwedge m \neq -\frac{3}{2} \bigwedge n > 1$, **then**

$$\int \left(c + d x^2\right)^m \text{ArcTanh}[a x]^n \, dx \longrightarrow$$

$$-\frac{n \left(c + d x^2\right)^{m+1} \text{ArcTanh}[a x]^{n-1}}{4 a c (m+1)^2} - \frac{x \left(c + d x^2\right)^{m+1} \text{ArcTanh}[a x]^n}{2 c (m+1)} +$$
$$\frac{2 m + 3}{2 c (m+1)} \int \left(c + d x^2\right)^{m+1} \text{ArcTanh}[a x]^n \, dx + \frac{n (n-1)}{4 (m+1)^2} \int \left(c + d x^2\right)^m \text{ArcTanh}[a x]^{n-2} \, dx$$

- **Program code:**

```
Int[(c_+d_.*x_^2)^m_*ArcTanh[a_.*x_]^n_,x_Symbol] :=
  -n*(c+d*x^2)^(m+1)*ArcTanh[a*x]^(n-1)/(4*a*c*(m+1)^2) -
  x*(c+d*x^2)^(m+1)*ArcTanh[a*x]^n/(2*c*(m+1)) +
  Dist[(2*m+3)/(2*c*(m+1)),Int[(c+d*x^2)^(m+1)*ArcTanh[a*x]^n,x]] +
  Dist[n*(n-1)/(4*(m+1)^2),Int[(c+d*x^2)^m*ArcTanh[a*x]^(n-2),x]] /;
FreeQ[{a,c,d},x] && ZeroQ[a^2*c+d] && RationalQ[{m,n}] && m<-1 && m≠-3/2 && n>1
```

- **Derivation: Integration by parts**
- **Rule: If** $a^2 c + d = 0 \wedge m < -1 \wedge n < -1$, **then**

$$\int \left(c + d x^2\right)^m \text{ArcTanh}[a x]^n \, dx \longrightarrow$$

$$\frac{\left(c + d x^2\right)^{m+1} \text{ArcTanh}[a x]^{n+1}}{a c (n+1)} + \frac{2 a (m+1)}{n+1} \int x \left(c + d x^2\right)^m \text{ArcTanh}[a x]^{n+1} \, dx$$

- **Program code:**

```
Int[(c_+d_.*x_^2)^m_*ArcTanh[a_.*x_]^n_,x_Symbol] :=
  (c+d*x^2)^(m+1)*ArcTanh[a*x]^(n+1)/(a*c*(n+1)) +
  Dist[2*a*(m+1)/(n+1),Int[x*(c+d*x^2)^m*ArcTanh[a*x]^(n+1),x]] /;
FreeQ[{a,c,d},x] && ZeroQ[a^2*c+d] && RationalQ[{m,n}] && m<-1 && n<-1
```

- **Derivation: Integration by substitution**
- **Basis:** $(1 - a^2 x^2)^m \text{ArcTanh}[a x]^n = \frac{1}{a} \text{Sech}[\text{ArcTanh}[a x]]^{2(m+1)} \text{ArcTanh}[a x]^n \partial_x \text{ArcTanh}[a x]$
- **Rule: If** $a^2 c + d = 0 \wedge m, n \in \mathbb{Q} \wedge m < -1 \wedge (n < 0 \vee n \notin \mathbb{Z}) \wedge (m \in \mathbb{Z} \vee c > 0)$, **then**

$$\int \left(c + d x^2\right)^m \text{ArcTanh}[a x]^n \, dx \longrightarrow \frac{c^m}{a} \text{Subst}\left[\int x^n \text{Sech}[x]^{2(m+1)} \, dx,\ x,\ \text{ArcTanh}[a x]\right]$$

- **Program code:**

```
Int[(c_+d_.*x_^2)^m_*ArcTanh[a_.*x_]^n_,x_Symbol] :=
  Dist[c^m/a,Subst[Int[x^n*Sech[x]^(2*(m+1)),x],x,ArcTanh[a*x]]] /;
FreeQ[{a,c,d},x] && ZeroQ[a^2*c+d] && RationalQ[{m,n}] && m<-1 && (n<0 || Not[IntegerQ[n]]) && (Inte
```

- **Basis: If** $a^2 c + d = 0$, $D\left[\frac{c^{m-\frac{1}{2}} \sqrt{c+d x^2}}{\sqrt{1-a^2 x^2}}, x\right] = 0$
- **Rule: If** $a^2 c + d = 0 \wedge m, n \in \mathbb{Q} \bigwedge m < -1 \bigwedge (n < 0 \vee n \notin \mathbb{Z}) \bigwedge m - \frac{1}{2} \in \mathbb{Z} \bigwedge \neg (c > 0)$, **then**

$$\int \left(c + d x^2\right)^m \text{ArcTanh}[a x]^n \, dx \longrightarrow \frac{c^{m-\frac{1}{2}} \sqrt{c + d x^2}}{\sqrt{1 - a^2 x^2}} \int \left(1 - a^2 x^2\right)^m \text{ArcTanh}[a x]^n \, dx$$

- **Program code:**

```
Int[(c_+d_.*x_^2)^m_*ArcTanh[a_.*x_]^n_,x_Symbol] :=
  c^(m-1/2)*Sqrt[c+d*x^2]/Sqrt[1-a^2*x^2]*Int[(1-a^2*x^2)^m*ArcTanh[a*x]^n,x] /;
FreeQ[{a,c,d},x] && ZeroQ[a^2*c+d] && RationalQ[{m,n}] && m<-1 && (n<0 || Not[IntegerQ[n]]) && Integ
```

$$\int x^m \left(c+d\,x^2\right)^p \text{ArcTanh}[a\,x]^n\,dx$$

- **Derivation: Integration by parts**

- **Rule: If** $a^2 c + d = 0 \land p \in \mathbb{Q} \land n > 0 \land p \neq -1$, **then**

$$\int x\left(c+d\,x^2\right)^p \text{ArcTanh}[a\,x]^n\,dx \longrightarrow \frac{\left(c+d\,x^2\right)^{p+1}\text{ArcTanh}[a\,x]^n}{2\,d\,(p+1)} + \frac{n}{2\,a\,(p+1)}\int\left(c+d\,x^2\right)^p \text{ArcTanh}[a\,x]^{n-1}\,dx$$

- **Program code:**

```
Int[x_*(c_+d_.*x_^2)^p_.*ArcTanh[a_.*x_]^n_.,x_Symbol] :=
  (c+d*x^2)^(p+1)*ArcTanh[a*x]^n/(2*d*(p+1)) +
  Dist[n/(2*a*(p+1)),Int[(c+d*x^2)^p*ArcTanh[a*x]^(n-1),x]] /;
FreeQ[{a,c,d},x] && ZeroQ[a^2*c+d] && RationalQ[{n,p}] && n>0 && p≠-1
```

- **Rule: If** $a^2 c + d = 0 \land p \in \mathbb{Q}$, **then**

$$\int \frac{x\left(c+d\,x^2\right)^p}{\text{ArcTanh}[a\,x]^2}\,dx \longrightarrow -\frac{x\left(c+d\,x^2\right)^{p+1}}{a\,c\,\text{ArcTanh}[a\,x]} + \frac{1}{a}\int\frac{\left(1-(2\,p+3)\,a^2\,x^2\right)\left(c+d\,x^2\right)^p}{\text{ArcTanh}[a\,x]}\,dx$$

- **Program code:**

```
Int[x_*(c_+d_.*x_^2)^p_./ArcTanh[a_.*x_]^2,x_Symbol] :=
  -x*(c+d*x^2)^(p+1)/(a*c*ArcTanh[a*x]) +
  Dist[1/a,Int[(1-(2*p+3)*a^2*x^2)*(c+d*x^2)^p/ArcTanh[a*x],x]] /;
FreeQ[{a,c,d},x] && ZeroQ[a^2*c+d] && RationalQ[p]
```

- **Rule: If** $a^2 c + d = 0 \land n < -1 \land n \neq -2$, **then**

$$\int \frac{x\,\text{ArcTanh}[a\,x]^n}{\left(c+d\,x^2\right)^2}\,dx \longrightarrow$$

$$\frac{x\,\text{ArcTanh}[a\,x]^{n+1}}{a\,c\,(n+1)\left(c+d\,x^2\right)} + \frac{\left(1+a^2\,x^2\right)\text{ArcTanh}[a\,x]^{n+2}}{d\,(n+1)\,(n+2)\left(c+d\,x^2\right)} + \frac{4}{(n+1)\,(n+2)}\int\frac{x\,\text{ArcTanh}[a\,x]^{n+2}}{\left(c+d\,x^2\right)^2}\,dx$$

- **Program code:**

```
Int[x_*ArcTanh[a_.*x_]^n_/(c_+d_.*x_^2)^2,x_Symbol] :=
  x*ArcTanh[a*x]^(n+1)/(a*c*(n+1)*(c+d*x^2)) +
  (1+a^2*x^2)*ArcTanh[a*x]^(n+2)/(d*(n+1)*(n+2)*(c+d*x^2)) +
  Dist[4/((n+1)*(n+2)),Int[x*ArcTanh[a*x]^(n+2)/(c+d*x^2)^2,x]] /;
FreeQ[{a,c,d},x] && ZeroQ[a^2*c+d] && RationalQ[n] && n<-1 && n≠-2
```

- **Derivation: Integration by parts**
- **Rule: If** $a^2 c + d = 0 \wedge m, n, 2p \in \mathbb{Z} \wedge m < -1 \wedge n > 0 \wedge m + 2p + 3 = 0$, **then**

$$\int x^m \left(c + d x^2\right)^p \text{ArcTanh}[a x]^n \, dx \longrightarrow$$

$$\frac{x^{m+1} \left(c + d x^2\right)^{p+1} \text{ArcTanh}[a x]^n}{c (m+1)} - \frac{a n}{m+1} \int x^{m+1} \left(c + d x^2\right)^p \text{ArcTanh}[a x]^{n-1} \, dx$$

- **Program code:**

```
Int[x_^m_*(c_+d_.*x_^2)^p_.*ArcTanh[a_.*x_]^n_.,x_Symbol] :=
  x^(m+1)*(c+d*x^2)^(p+1)*ArcTanh[a*x]^n/(c*(m+1)) -
  Dist[a*n/(m+1),Int[x^(m+1)*(c+d*x^2)^p*ArcTanh[a*x]^(n-1),x]] /;
FreeQ[{a,c,d},x] && ZeroQ[a^2*c+d] && IntegersQ[m,n,2*p] && m<-1 && n>0 && ZeroQ[m+2*p+3]
```

- **Derivation: Integration by parts**
- **Rule: If** $a^2 c + d = 0 \wedge m, n, 2p \in \mathbb{Z} \wedge n < -1 \wedge m + 2p + 2 = 0$, **then**

$$\int x^m \left(c + d x^2\right)^p \text{ArcTanh}[a x]^n \, dx \longrightarrow$$

$$\frac{x^m \left(c + d x^2\right)^{p+1} \text{ArcTanh}[a x]^{n+1}}{a c (n+1)} - \frac{m}{a (n+1)} \int x^{m-1} \left(c + d x^2\right)^p \text{ArcTanh}[a x]^{n+1} \, dx$$

- **Program code:**

```
Int[x_^m_*(c_+d_.*x_^2)^p_.*ArcTanh[a_.*x_]^n_.,x_Symbol] :=
  x^m*(c+d*x^2)^(p+1)*ArcTanh[a*x]^(n+1)/(a*c*(n+1)) -
  Dist[m/(a*(n+1)),Int[x^(m-1)*(c+d*x^2)^p*ArcTanh[a*x]^(n+1),x]] /;
FreeQ[{a,c,d},x] && ZeroQ[a^2*c+d] && IntegersQ[m,n,2*p] && n<-1 && ZeroQ[m+2*p+2]
```

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{x^2}{c+d\,x^2} = \frac{1}{d} - \frac{c}{d\,(c+d\,x^2)}$
- **Rule: If** $a^2\,c + d = 0\,\bigwedge\,m,\,n,\,2\,p \in \mathbb{Z}\,\bigwedge\,m > 1\,\bigwedge\,n \neq -1\,\bigwedge\,p < -1$**, then**

$$\int x^m\,\left(c + d\,x^2\right)^p\,\text{ArcTanh}[a\,x]^n\,dx \,\longrightarrow\, \frac{1}{d}\int x^{m-2}\,\left(c + d\,x^2\right)^{p+1}\,\text{ArcTanh}[a\,x]^n\,dx - \frac{c}{d}\int x^{m-2}\,\left(c + d\,x^2\right)^p\,\text{ArcTanh}[a\,x]^n\,dx$$

- **Program code:**

```
Int[x_^m_*(c_+d_.*x_^2)^p_*ArcTanh[a_.*x_]^n_.,x_Symbol] :=
  Dist[1/d,Int[x^(m-2)*(c+d*x^2)^(p+1)*ArcTanh[a*x]^n,x]] -
  Dist[c/d,Int[x^(m-2)*(c+d*x^2)^p*ArcTanh[a*x]^n,x]] /;
FreeQ[{a,c,d},x] && ZeroQ[a^2*c+d] && IntegersQ[m,n,2*p] && m>1 && n≠-1 && p<-1
```

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{1}{c+d\,x^2} = \frac{1}{c} - \frac{d\,x^2}{c\,(c+d\,x^2)}$
- **Rule: If** $a^2\,c + d = 0\,\bigwedge\,m,\,n,\,2\,p \in \mathbb{Z}\,\bigwedge\,m < 0\,\bigwedge\,n \neq -1\,\bigwedge\,p < -1$**, then**

$$\int x^m\,\left(c + d\,x^2\right)^p\,\text{ArcTanh}[a\,x]^n\,dx \,\longrightarrow\, \frac{1}{c}\int x^m\,\left(c + d\,x^2\right)^{p+1}\,\text{ArcTanh}[a\,x]^n\,dx - \frac{d}{c}\int x^{m+2}\,\left(c + d\,x^2\right)^p\,\text{ArcTanh}[a\,x]^n\,dx$$

- **Program code:**

```
Int[x_^m_*(c_+d_.*x_^2)^p_*ArcTanh[a_.*x_]^n_.,x_Symbol] :=
  Dist[1/c,Int[x^m*(c+d*x^2)^(p+1)*ArcTanh[a*x]^n,x]] -
  Dist[d/c,Int[x^(m+2)*(c+d*x^2)^p*ArcTanh[a*x]^n,x]] /;
FreeQ[{a,c,d},x] && ZeroQ[a^2*c+d] && IntegersQ[m,n,2*p] && m<0 && n≠-1 && p<-1
```

- **Derivation: Integration by parts**
- **Rule: If** $a^2 c + d = 0 \wedge m, n, 2p \in \mathbb{Z} \wedge m < -1 \wedge n > 0 \wedge m + 2p + 3 \neq 0$, **then**

$$\int x^m \left(c + d x^2\right)^p \text{ArcTanh}[a x]^n \, dx \longrightarrow \frac{x^{m+1} \left(c + d x^2\right)^{p+1} \text{ArcTanh}[a x]^n}{c (m + 1)} - \frac{a n}{m + 1} \int x^{m+1} \left(c + d x^2\right)^p \text{ArcTanh}[a x]^{n-1} \, dx + \frac{a^2 (m + 2p + 3)}{m + 1} \int x^{m+2} \left(c + d x^2\right)^p \text{ArcTanh}[a x]^n \, dx$$

- **Program code:**

```
Int[x_^m_*(c_+d_.*x_^2)^p_.*ArcTanh[a_.*x_]^n_.,x_Symbol] :=
  x^(m+1)*(c+d*x^2)^(p+1)*ArcTanh[a*x]^n/(c*(m+1)) -
  Dist[a*n/(m+1),Int[x^(m+1)*(c+d*x^2)^p*ArcTanh[a*x]^(n-1),x]] +
  Dist[a^2*(m+2*p+3)/(m+1),Int[x^(m+2)*(c+d*x^2)^p*ArcTanh[a*x]^n,x]] /;
FreeQ[{a,c,d},x] && ZeroQ[a^2*c+d] && IntegersQ[m,n,2*p] && m<-1 && n>0 && NonzeroQ[m+2*p+3]
```

- **Derivation: Integration by parts**
- **Rule: If** $a^2 c + d = 0 \wedge m, n, 2p \in \mathbb{Z} \wedge n < -1 \wedge m + 2p + 2 \neq 0$, **then**

$$\int x^m \left(c + d x^2\right)^p \text{ArcTanh}[a x]^n \, dx \longrightarrow \frac{x^m \left(c + d x^2\right)^{p+1} \text{ArcTanh}[a x]^{n+1}}{a c (n + 1)} - \frac{m}{a (n + 1)} \int x^{m-1} \left(c + d x^2\right)^p \text{ArcTanh}[a x]^{n+1} \, dx + \frac{a (m + 2p + 2)}{n + 1} \int x^{m+1} \left(c + d x^2\right)^p \text{ArcTanh}[a x]^{n+1} \, dx$$

- **Program code:**

```
Int[x_^m_.*(c_+d_.*x_^2)^p_.*ArcTanh[a_.*x_]^n_.,x_Symbol] :=
  x^m*(c+d*x^2)^(p+1)*ArcTanh[a*x]^(n+1)/(a*c*(n+1)) -
  Dist[m/(a*(n+1)),Int[x^(m-1)*(c+d*x^2)^p*ArcTanh[a*x]^(n+1),x]] +
  Dist[a*(m+2*p+2)/(n+1),Int[x^(m+1)*(c+d*x^2)^p*ArcTanh[a*x]^(n+1),x]] /;
FreeQ[{a,c,d},x] && ZeroQ[a^2*c+d] && IntegersQ[m,n,2*p] && n<-1 && NonzeroQ[m+2*p+2] && Not[m==1 &&
```

- **Derivation: Integration by substitution**

- **Basis: If** $m \in \mathbb{Z}$ **or** $a > 0$, $(e + f x^m) (1 - a^2 x^2)^p \text{ArcTanh}[a x]^n =$
$\frac{1}{a^{m+1}} (e a^m + f \text{Tanh}[\text{ArcTanh}[a x]]^m) \text{Sech}[\text{ArcTanh}[a x]]^{2 (p+1)} \text{ArcTanh}[a x]^n \partial_x \text{ArcTanh}[a x]$

- **Rule: If** $a^2 c + d = 0 \land m, n, p \in \mathbb{Q} \land p < -1 \land (n < 0 \lor n \notin \mathbb{Z}) \land (p \in \mathbb{Z} \lor c > 0) \land (m \in \mathbb{Z} \lor a > 0)$**, then**

$$\int (e + f x^m) \left(c + d x^2\right)^p \text{ArcTanh}[a x]^n \, dx \longrightarrow \frac{c^p}{a^{m+1}}$$
$$\text{Subst}\left[\int x^n (e a^m + f \text{Tanh}[x]^m) \text{Sech}[x]^{2 (p+1)} \, dx, x, \text{ArcTanh}[a x]\right]$$

- **Program code:**

```
Int[(e_.+f_.*x_^m_.)*(c_+d_.*x_^2)^p_*ArcTanh[a_.*x_]^n_,x_Symbol] :=
  Dist[c^p/a^(m+1),Subst[Int[Expand[x^n*TrigReduce[Regularize[(e*a^m+f*Tanh[x]^m)*Sech[x]^(2*(p+1)),:
FreeQ[{a,c,d,e,f},x] && ZeroQ[a^2*c+d] && RationalQ[{m,n,p}] && p<-1 && (n<0 || Not[IntegerQ[n]]) &&
```

- **Derivation: Integration by substitution**

- **Basis:** $x^m (1 - a^2 x^2)^p \text{ArcTanh}[a x]^n = \frac{1}{a} \left(\frac{\text{Tanh}[\text{ArcTanh}[a x]]}{a}\right)^m \text{Sech}[\text{ArcTanh}[a x]]^{2 (p+1)} \text{ArcTanh}[a x]^n \partial_x \text{ArcTanh}[a x]$

- **Rule: If** $a^2 c + d = 0 \land m, n \in \mathbb{Q} \land p < -1 \land (n < 0 \lor n \notin \mathbb{Z}) \land (p \in \mathbb{Z} \lor c > 0) \land \neg (m \in \mathbb{Z} \lor a > 0)$**, then**

$$\int x^m \left(c + d x^2\right)^p \text{ArcTanh}[a x]^n \, dx \longrightarrow \frac{c^p}{a} \text{Subst}\left[\int x^n (\text{Tanh}[x] / a)^m \text{Sech}[x]^{2 (p+1)} \, dx, x, \text{ArcTanh}[a x]\right]$$

- **Program code:**

```
Int[x_^m_.*(c_+d_.*x_^2)^p_*ArcTanh[a_.*x_]^n_,x_Symbol] :=
  Dist[c^p/a,Subst[Int[x^n*(Tanh[x]/a)^m*Sech[x]^(2*(p+1)),x],x,ArcTanh[a*x]]] /;
FreeQ[{a,c,d},x] && ZeroQ[a^2*c+d] && RationalQ[{m,n,p}] && p<-1 && (n<0 || Not[IntegerQ[n]]) && (In
```

- **Basis: If** $a^2 c + d = 0$, $D\left[\frac{c^{p-\frac{1}{2}} \sqrt{c+d x^2}}{\sqrt{1-a^2 x^2}}, x\right] = 0$

- **Rule: If** $a^2 c + d = 0 \land m, n \in \mathbb{Q} \bigwedge p < -1 \bigwedge (n < 0 \lor n \notin \mathbb{Z}) \bigwedge p - \frac{1}{2} \in \mathbb{Z} \bigwedge \neg (c > 0)$**, then**

$$\int x^m \left(c + d x^2\right)^p \text{ArcTanh}[a x]^n \, dx \longrightarrow \frac{c^{p-\frac{1}{2}} \sqrt{c + d x^2}}{\sqrt{1 - a^2 x^2}} \int x^m \left(1 - a^2 x^2\right)^p \text{ArcTanh}[a x]^n \, dx$$

- **Program code:**

```
Int[x_^m_.*(c_+d_.*x_^2)^p_*ArcTanh[a_.*x_]^n_,x_Symbol] :=
  c^(p-1/2)*Sqrt[c+d*x^2]/Sqrt[1-a^2*x^2]*Int[x^m*(1-a^2*x^2)^p*ArcTanh[a*x]^n,x] /;
FreeQ[{a,c,d},x] && ZeroQ[a^2*c+d] && RationalQ[{m,n,p}] && p<-1 && (n<0 || Not[IntegerQ[n]]) && Int
```

$\int \text{ArcTanh}[a + b\, x^n]\, dx$

- **Reference: CRC 585, A&S 4.6.45**
- **Derivation: Integration by parts**
- **Rule:**

$$\int \text{ArcTanh}[a + b\, x]\, dx \longrightarrow \frac{(a + b\, x)\, \text{ArcTanh}[a + b\, x]}{b} + \frac{\text{Log}\left[1 - (a + b\, x)^2\right]}{2\, b}$$

- **Program code:**

```
Int[ArcTanh[a_+b_.*x_],x_Symbol] :=
  (a+b*x)*ArcTanh[a+b*x]/b + Log[1-(a+b*x)^2]/(2*b) /;
FreeQ[{a,b},x]
```

- **Reference: CRC 585, A&S 4.6.45**
- **Derivation: Integration by parts**
- **Rule: If $n \in \mathbb{Q}$, then**

$$\int \text{ArcTanh}[a + b\, x^n]\, dx \longrightarrow x\, \text{ArcTanh}[a + b\, x^n] - b\, n \int \frac{x^n}{1 - a^2 - 2\, a\, b\, x^n - b^2\, x^{2\, n}}\, dx$$

- **Program code:**

```
Int[ArcTanh[a_+b_.*x_^n_],x_Symbol] :=
  x*ArcTanh[a+b*x^n] -
  Dist[b*n,Int[x^n/(1-a^2-2*a*b*x^n-b^2*x^(2*n)),x]] /;
FreeQ[{a,b},x] && RationalQ[n]
```

$$\int x^m \operatorname{ArcTanh}[a + b x^n] \, dx$$

- **Derivation: Algebraic expansion**
- **Basis:** $\operatorname{ArcTanh}[z] = \frac{1}{2} \operatorname{Log}[1+z] - \frac{1}{2} \operatorname{Log}[1-z]$
- **Rule:**

$$\int \frac{\operatorname{ArcTanh}[a + b x^n]}{x} \, dx \longrightarrow \frac{1}{2} \int \frac{\operatorname{Log}[1 + a + b x^n]}{x} \, dx - \frac{1}{2} \int \frac{\operatorname{Log}[1 - a - b x^n]}{x} \, dx$$

- **Program code:**

```
Int[ArcTanh[a_.+b_.*x_^n_.]/x_,x_Symbol] :=
  Dist[1/2,Int[Log[1+a+b*x^n]/x,x]] -
  Dist[1/2,Int[Log[1-a-b*x^n]/x,x]] /;
FreeQ[{a,b,n},x]
```

- **Reference: CRC 588, A&S 4.6.54**
- **Derivation: Integration by parts**
- **Rule: If** $m, n \in \mathbb{Q} \land m+1 \neq 0 \land m+1 \neq n$**, then**

$$\int x^m \operatorname{ArcTanh}[a + b x^n] \, dx \longrightarrow \frac{x^{m+1} \operatorname{ArcTanh}[a + b x^n]}{m+1} - \frac{b\, n}{m+1} \int \frac{x^{m+n}}{1 - a^2 - 2 a b x^n - b^2 x^{2n}} \, dx$$

- **Program code:**

```
Int[x_^m_.*ArcTanh[a_+b_.*x_^n_.],x_Symbol] :=
  x^(m+1)*ArcTanh[a+b*x^n]/(m+1) -
  Dist[b*n/(m+1),Int[x^(m+n)/(1-a^2-2*a*b*x^n-b^2*x^(2*n)),x]] /;
FreeQ[{a,b},x] && RationalQ[{m,n}] && m+1≠0 && m+1≠n
```

$$\int \text{ArcTanh}[a+b\,x]^n\,dx$$

- **Derivation: Integration by substitution**
- **Rule: If $n \in \mathbb{Z} \land n > 1$, then**

$$\int \text{ArcTanh}[a+b\,x]^n\,dx \longrightarrow \frac{1}{b}\,\text{Subst}\Big[\int \text{ArcTanh}[x]^n\,dx,\,x,\,a+b\,x\Big]$$

- **Program code:**

```
Int[ArcTanh[a_+b_.*x_]^n_,x_Symbol] :=
  Dist[1/b,Subst[Int[ArcTanh[x]^n,x],x,a+b*x]] /;
FreeQ[{a,b},x] && IntegerQ[n] && n>1
```

$$\int x^m \operatorname{ArcTanh}[a+b\,x]^n \, dx$$

- **Derivation: Integration by substitution**
- **Rule: If $m, n \in \mathbb{Z} \wedge m > 0 \wedge n > 1$, then**

$$\int x^m \operatorname{ArcTanh}[a+b\,x]^n \, dx \longrightarrow \frac{1}{b^{m+1}} \operatorname{Subst}\left[\int (x-a)^m \operatorname{ArcTanh}[x]^n \, dx,\ x,\ a+b\,x\right]$$

- **Program code:**

```
Int[x_^m_.*ArcTanh[a_+b_.*x_]^n_,x_Symbol] :=
  Dist[1/b^(m+1),Subst[Int[(x-a)^m*ArcTanh[x]^n,x],x,a+b*x]] /;
FreeQ[{a,b},x] && IntegersQ[m,n] && m>0 && n>1
```

$$\int \frac{\text{ArcTanh}[a + b\,x]}{c + d\,x^n}\,dx$$

- **Derivation: Algebraic simplification**
- **Basis:** $\text{ArcTanh}[z] = \frac{1}{2}\,\text{Log}[1 + z] - \frac{1}{2}\,\text{Log}[1 - z]$
- **Rule: If** $n \in \mathbb{Z} \land \neg\,(n = 2 \land b^2\,c + d = 0)$**, then**

$$\int \frac{\text{ArcTanh}[b\,x]}{c + d\,x^n}\,dx \longrightarrow \frac{1}{2}\int \frac{\text{Log}[1 + b\,x]}{c + d\,x^n}\,dx - \frac{1}{2}\int \frac{\text{Log}[1 - b\,x]}{c + d\,x^n}\,dx$$

- **Program code:**

```
Int[ArcTanh[b_.*x_]/(c_+d_.*x_^n_.),x_Symbol] :=
  Dist[1/2,Int[Log[1+b*x]/(c+d*x^n),x]] -
  Dist[1/2,Int[Log[1-b*x]/(c+d*x^n),x]] /;
FreeQ[{b,c,d},x] && IntegerQ[n] && Not[n==2 && ZeroQ[b^2*c+d]]
```

- **Derivation: Algebraic simplification**
- **Basis:** $\text{ArcTanh}[z] = \frac{1}{2}\,\text{Log}[1 + z] - \frac{1}{2}\,\text{Log}[1 - z]$
- **Rule: If** $n \in \mathbb{Z} \land \neg\,(n = 1 \land a\,d - b\,c = 0)$**, then**

$$\int \frac{\text{ArcTanh}[a + b\,x]}{c + d\,x^n}\,dx \longrightarrow \frac{1}{2}\int \frac{\text{Log}[1 + a + b\,x]}{c + d\,x^n}\,dx - \frac{1}{2}\int \frac{\text{Log}[1 - a - b\,x]}{c + d\,x^n}\,dx$$

- **Program code:**

```
Int[ArcTanh[a_+b_.*x_]/(c_+d_.*x_^n_.),x_Symbol] :=
  Dist[1/2,Int[Log[1+a+b*x]/(c+d*x^n),x]] -
  Dist[1/2,Int[Log[1-a-b*x]/(c+d*x^n),x]] /;
FreeQ[{a,b,c,d},x] && IntegerQ[n] && Not[n==1 && ZeroQ[a*d-b*c]]
```

$$\int u \,\text{ArcTanh}\left[\frac{c}{a+b\,x^n}\right]^m dx$$

- **Derivation: Algebraic simplification**
- **Basis:** $\text{ArcTanh}[z] = \text{ArcCoth}\left[\frac{1}{z}\right]$
- **Rule:**

$$\int u \,\text{ArcTanh}\left[\frac{c}{a+b\,x^n}\right]^m dx \longrightarrow \int u \,\text{ArcCoth}\left[\frac{a}{c}+\frac{b\,x^n}{c}\right]^m dx$$

- **Program code:**

```
Int[u_.*ArcTanh[c_./(a_.+b_.*x_^n_.)]^m_.,x_Symbol] :=
  Int[u*ArcCoth[a/c+b*x^n/c]^m,x] /;
FreeQ[{a,b,c,n,m},x]
```

$$\int \frac{f[x, \text{ArcTanh}[a + b x]]}{1 - (a + b x)^2} \, dx$$

- **Derivation: Integration by substitution**
- **Basis:** $\frac{f[z]}{1-z^2} = f[\text{Tanh}[\text{ArcTanh}[z]]] \, \text{ArcTanh}'[z]$
- **Basis:** $r + s x + t x^2 = -\frac{s^2-4 r t}{4 t} \left(1 - \frac{(s+2 t x)^2}{s^2-4 r t}\right)$
- **Basis:** $1 - \text{Tanh}[z]^2 = \text{Sech}[z]^2$
- **Rule:**

$$\int \frac{f[x, \text{ArcTanh}[a + b x]]}{1 - (a + b x)^2} \, dx \longrightarrow \frac{1}{b} \, \text{Subst}\left[\int f\left[-\frac{a}{b} + \frac{\text{Tanh}[x]}{b}, x\right] dx, x, \text{ArcTanh}[a + b x]\right]$$

- **Program code:**

```
If[ShowSteps,

Int[u_*v_^n_.,x_Symbol] :=
  Module[{tmp=InverseFunctionOfLinear[u,x]},
  ShowStep["","Int[f[x,ArcTanh[a+b*x]]/(1-(a+b*x)^2),x]",
           "Subst[Int[f[-a/b+Tanh[x]/b,x],x],x,ArcTanh[a+b*x]]/b",Hold[
  Dist[(-Discriminant[v,x]/(4*Coefficient[v,x,2]))^n/Coefficient[tmp[[1]],x,1],
    Subst[Int[Regularize[SubstForInverseFunction[u,tmp,x]*Sech[x]^(2*(n+1)),x],x], x, tmp]]]] /;
 NotFalseQ[tmp] && Head[tmp]===ArcTanh && ZeroQ[Discriminant[v,x]*tmp[[1]]^2-D[v,x]^2]] /;
SimplifyFlag && QuadraticQ[v,x] && IntegerQ[n] && n<0 && PosQ[Discriminant[v,x]] && MatchQ[u,r_.*f_^

Int[u_*v_^n_.,x_Symbol] :=
  Module[{tmp=InverseFunctionOfLinear[u,x]},
  Dist[(-Discriminant[v,x]/(4*Coefficient[v,x,2]))^n/Coefficient[tmp[[1]],x,1],
    Subst[Int[Regularize[SubstForInverseFunction[u,tmp,x]*Sech[x]^(2*(n+1)),x],x], x, tmp]] /;
 NotFalseQ[tmp] && Head[tmp]===ArcTanh && ZeroQ[Discriminant[v,x]*tmp[[1]]^2-D[v,x]^2]] /;
QuadraticQ[v,x] && IntegerQ[n] && n<0 && PosQ[Discriminant[v,x]] && MatchQ[u,r_.*f_^w_ /; FreeQ[f,x]
```

$$\int u\, e^{n\,\text{ArcTanh}[v]}\, dx$$

- **Derivation: Algebraic simplification**
- **Basis:** $e^{n\,\text{ArcTanh}[z]} = \frac{(1+z)^{n/2}}{(1-z)^{n/2}}$
- **Rule: If** $\frac{n}{2} \in \mathbb{Z}$**, then**

$$\int u\, e^{n\,\text{ArcTanh}[v]}\, dx \longrightarrow \int \frac{u\,(1+v)^{n/2}}{(1-v)^{n/2}}\, dx$$

- **Program code:**

```
Int[u_.*E^(n_.*ArcTanh[v_]),x_Symbol] :=
  Int[u*(1+v)^(n/2)/(1-v)^(n/2),x] /;
EvenQ[n]
```

- **Derivation: Algebraic simplification**
- **Basis:** $e^{n\,\text{ArcTanh}[z]} = \frac{(1+z)^{n/2}}{(1-z)^{n/2}}$
- **Rule: If** $n \in \mathbb{Q}$**, then**

$$\int e^{n\,\text{ArcTanh}[v]}\, dx \longrightarrow \int \frac{(1+v)^{n/2}}{(1-v)^{n/2}}\, dx$$

- **Program code:**

```
Int[E^(n_.*ArcTanh[v_]),x_Symbol] :=
  Int[(1+v)^(n/2)/(1-v)^(n/2),x] /;
RationalQ[n]
```

- **Derivation: Algebraic simplification**
- **Basis:** $e^{n\,\text{ArcTanh}[z]} = \left(\frac{1+z}{\sqrt{1-z^2}}\right)^n$
- **Rule: If** $m \in \mathbb{Q} \bigwedge \frac{n-1}{2} \in \mathbb{Z} \bigwedge$ **v is a polynomial, then**

$$\int x^m e^{n\,\text{ArcTanh}[v]}\,dx \longrightarrow \int \frac{x^m (1+v)^n}{\left(1-v^2\right)^{n/2}}\,dx$$

- **Program code:**

```
Int[x_^m_.*E^(n_.*ArcTanh[v_]), x_Symbol] :=
  Int[x^m*(1+v)^n/(1-v^2)^(n/2),x] /;
RationalQ[m] && OddQ[n] && PolynomialQ[v,x]
```

- **Derivation: Algebraic simplification**
- **Basis:** $e^{n\,\text{ArcTanh}[z]} (1-z^2)^m = (1-z)^{m-\frac{n}{2}} (1+z)^{m+\frac{n}{2}}$
- **Rule: If** $m,\ n \in \mathbb{Q} \bigwedge m-\frac{n}{2} \in \mathbb{Z} \bigwedge m+\frac{n}{2} \in \mathbb{Z}$**, then**

$$\int u\, e^{n\,\text{ArcTanh}[v]} \left(1-v^2\right)^m dx \longrightarrow \int u\,(1-v)^{m-\frac{n}{2}} (1+v)^{m+\frac{n}{2}}\,dx$$

- **Program code:**

```
Int[u_.*E^(n_.*ArcTanh[v_])*(1-v_^2)^m_.,x_Symbol] :=
  Int[u*(1-v)^(m-n/2)*(1+v)^(m+n/2),x] /;
RationalQ[{m,n}] && IntegerQ[m-n/2] && IntegerQ[m+n/2]
```

- **Derivation: Algebraic simplification**
- **Basis:** $e^{n\,\text{ArcTanh}[z]} (1-z^2)^m = (1-z)^{m-\frac{n}{2}} (1+z)^{m+\frac{n}{2}}$
- **Rule: If** $m,\ n \in \mathbb{Q} \bigwedge m-\frac{n}{2} \in \mathbb{Z} \bigwedge m+\frac{n}{2} \in \mathbb{Z}$**, then**

$$\int u\, e^{n\,\text{ArcTanh}[v]} \left(a-a\,v^2\right)^m dx \longrightarrow \frac{\left(a-a\,v^2\right)^m}{\left(1-v^2\right)^m} \int u\,(1-v)^{m-\frac{n}{2}} (1+v)^{m+\frac{n}{2}}\,dx$$

- **Program code:**

```
Int[u_.*E^(n_.*ArcTanh[v_])*(a_+b_.*v_^2)^m_.,x_Symbol] :=
  (a+b*v^2)^m/(1-v^2)^m*Int[u*(1-v)^(m-n/2)*(1+v)^(m+n/2),x] /;
FreeQ[{a,b},x] && ZeroQ[a+b] && RationalQ[{m,n}] && IntegerQ[m-n/2] && IntegerQ[m+n/2]
```

- **Derivation: Algebraic simplification**
- **Basis:** $e^{n\,\mathrm{ArcTanh}[z]} = \frac{(1+z)^n}{(1-z^2)^{n/2}}$
- **Rule: If** $n \in \mathbb{Q} \wedge m \in \mathbb{Z} \wedge m > 0$**, then**

$$\int u\, e^{n\,\mathrm{ArcTanh}[v]} \left(a - a v^2\right)^m dx \longrightarrow a^m \int u\,(1+v)^n \left(1-v^2\right)^{m-\frac{n}{2}} dx$$

- **Program code:**

```
Int[u_.*E^(n_.*ArcTanh[v_])*(a_+b_.v_^2)^m_.,x_Symbol] :=
  Dist[a^m,Int[u*(1+v)^n*(1-v^2)^(m-n/2),x]] /;
FreeQ[{a,b},x] && ZeroQ[a+b] && RationalQ[n] && IntegerQ[m] && m>0
```

- **Derivation: Algebraic simplification**
- **Basis:** $e^{n\,\mathrm{ArcTanh}[z]} = \frac{(1+z)^n}{(1-z^2)^{n/2}}$
- **Rule: If** $m,\ n \in \mathbb{Q} \wedge m + n \in \mathbb{Z}$**, then**

$$\int u\, e^{n\,\mathrm{ArcTanh}[v]} (1+v)^m dx \longrightarrow \int \frac{u\,(1+v)^{m+n}}{\left(1-v^2\right)^{n/2}} dx$$

- **Program code:**

```
Int[u_.*E^(n_.*ArcTanh[v_])*(1+v_)^m_.,x_Symbol] :=
  Int[u*(1+v)^(m+n)/(1-v^2)^(n/2),x] /;
RationalQ[{m,n}] && IntegerQ[m+n]
```

- **Derivation: Algebraic simplification**
- **Basis:** $e^{n\,\mathrm{ArcTanh}[z]} = \frac{(1+z)^{n/2}}{(1-z)^{n/2}}$
- **Rule: If** $m,\ n \in \mathbb{Q}$**, then**

$$\int u\, e^{n\,\mathrm{ArcTanh}[v]} (1+v)^m dx \longrightarrow \int \frac{u\,(1+v)^{m+\frac{n}{2}}}{(1-v)^{n/2}} dx$$

- **Program code:**

```
Int[u_.*E^(n_.*ArcTanh[v_])*(1+v_)^m_.,x_Symbol] :=
  Int[u*(1+v)^(m+n/2)/(1-v)^(n/2),x] /;
RationalQ[{m,n}]
```

- **Derivation: Algebraic simplification**
- **Basis:** $e^{n\,\text{ArcTanh}[z]} = \frac{(1+z)^{n/2}}{(1-z)^{n/2}}$
- **Rule: If** $m, n \in \mathbb{Q}$**, then**

$$\int u\, e^{n\,\text{ArcTanh}[v]}\,(1-v)^m\,dx \longrightarrow \int u\,(1+v)^{n/2}\,(1-v)^{m-\frac{n}{2}}\,dx$$

- **Program code:**

```
Int[u_.*E^(n_.*ArcTanh[v_])*(1-v_)^m_.,x_Symbol] :=
  Int[u*(1+v)^(n/2)*(1-v)^(m-n/2),x] /;
RationalQ[{m,n}]
```

- **Derivation: Algebraic simplification**
- **Rule: If** $m \in \mathbb{Z} \land n \in \mathbb{Q} \land a-1 \neq 0 \land a^2-b^2=0$**, then**

$$\int u\, e^{n\,\text{ArcTanh}[v]}\,(a+b\,v)^m\,dx \longrightarrow a^m \int u\, e^{n\,\text{ArcTanh}[v]}\left(1+\frac{b\,v}{a}\right)^m dx$$

- **Program code:**

```
Int[u_.*E^(n_.*ArcTanh[v_])*(a_+b_.*v_)^m_.,x_Symbol] :=
  Dist[a^m,Int[u*E^(n*ArcTanh[v])*(1+b/a*v)^m,x]] /;
FreeQ[{a,b},x] && IntegerQ[m] && RationalQ[n] && NonzeroQ[a-1] && ZeroQ[a^2-b^2]
```

- **Derivation: Algebraic simplification**
- **Basis: If m is an integer,** $e^{\text{ArcTanh}[z]}\left(a-\frac{a}{z^2}\right)^m = \frac{(-a)^m\,(1+z)\,\left(1-z^2\right)^{m-\frac{1}{2}}}{z^{2\,m}}$
- **Rule: If** $m \in \mathbb{Z}$**, then**

$$\int u\, e^{\text{ArcTanh}[v]}\left(a-\frac{a}{v^2}\right)^m dx \longrightarrow (-a)^m \int \frac{u\,\left(1-v^2\right)^{m-\frac{1}{2}}}{v^{2\,m}}\,dx + (-a)^m \int \frac{u\,\left(1-v^2\right)^{m-\frac{1}{2}}}{v^{2\,m-1}}\,dx$$

- **Program code:**

```
Int[u_.*E^ArcTanh[v_]*(a_+b_./v_^2)^m_.,x_Symbol] :=
  b^m*Int[u*(1-v^2)^(m-1/2)/v^(2*m),x] +
  b^m*Int[u*(1-v^2)^(m-1/2)/v^(2*m-1),x] /;
FreeQ[{a,b},x] && ZeroQ[a+b] && IntegerQ[m]
```

$$\int \mathrm{ArcTanh}\left[b\, f^{c+d\,x}\right] dx$$

- **Derivation: Algebraic simplification**
- **Basis:** $\mathrm{ArcTanh}[z] = \frac{1}{2}\,\mathrm{Log}[1+z] - \frac{1}{2}\,\mathrm{Log}[1-z]$
- **Rule:**

$$\int \mathrm{ArcTanh}\left[b\, f^{c+d\,x}\right] dx \;\longrightarrow\; \frac{1}{2}\int \mathrm{Log}\left[1+b\, f^{c+d\,x}\right] dx - \frac{1}{2}\int \mathrm{Log}\left[1-b\, f^{c+d\,x}\right] dx$$

- **Program code:**

```
Int[ArcTanh[b_.*f_^(c_.+d_.*x_)],x_Symbol] :=
  Dist[1/2,Int[Log[1+b*f^(c+d*x)],x]] -
  Dist[1/2,Int[Log[1-b*f^(c+d*x)],x]] /;
FreeQ[{b,c,d,f},x]
```

$$\int x^m \,\text{ArcTanh}\left[b\, f^{c+d\,x}\right] dx$$

- **Derivation: Algebraic simplification**
- **Basis:** $\text{ArcTanh}[z] = \frac{1}{2}\,\text{Log}[1+z] - \frac{1}{2}\,\text{Log}[1-z]$
- **Rule: If** $m \in \mathbb{Z} \land m > 0$**, then**

$$\int x^m \,\text{ArcTanh}\left[b\, f^{c+d\,x}\right] dx \;\longrightarrow\; \frac{1}{2}\int x^m \,\text{Log}\left[1+b\, f^{c+d\,x}\right] dx - \frac{1}{2}\int x^m \,\text{Log}\left[1-b\, f^{c+d\,x}\right] dx$$

- **Program code:**

```
Int[x_^m_.*ArcTanh[b_.*f_^(c_.+d_.*x_)],x_Symbol] :=
  Dist[1/2,Int[x^m*Log[1+b*f^(c+d*x)],x]] -
  Dist[1/2,Int[x^m*Log[1-b*f^(c+d*x)],x]] /;
FreeQ[{b,c,d,f},x] && IntegerQ[m] && m>0
```

$$\int v\,\text{ArcTanh}[u]\,dx$$

- **Derivation: Integration by parts**
- **Rule: If u is free of inverse functions, then**

$$\int \text{ArcTanh}[u]\,dx \longrightarrow x\,\text{ArcTanh}[u] - \int \frac{x\,\partial_x u}{1-u^2}\,dx$$

- **Program code:**

```
Int[ArcTanh[u_],x_Symbol] :=
  x*ArcTanh[u] -
  Int[Regularize[x*D[u,x]/(1-u^2),x],x] /;
InverseFunctionFreeQ[u,x]
```

- **Derivation: Integration by parts**
- **Rule: If $m+1 \neq 0 \land$ u is free of inverse functions, then**

$$\int x^m\,\text{ArcTanh}[u]\,dx \longrightarrow \frac{x^{m+1}\,\text{ArcTanh}[u]}{m+1} - \frac{1}{m+1}\int \frac{x^{m+1}\,\partial_x u}{1-u^2}\,dx$$

- **Program code:**

```
Int[x_^m_.*ArcTanh[u_],x_Symbol] :=
  x^(m+1)*ArcTanh[u]/(m+1) -
  Dist[1/(m+1),Int[Regularize[x^(m+1)*D[u,x]/(1-u^2),x],x]] /;
FreeQ[m,x] && NonzeroQ[m+1] && InverseFunctionFreeQ[u,x] &&
    Not[FunctionOfQ[x^(m+1),u,x]] &&
    FalseQ[PowerVariableExpn[u,m+1,x]]
```

- **Derivation: Integration by parts**

- **Rule: If u is free of inverse functions, let $w = \int v \, dx$, if w is free of inverse functions, then**

$$\int v \operatorname{ArcTanh}[u] \, dx \longrightarrow w \operatorname{ArcTanh}[u] - \int \frac{w \, \partial_x u}{1 - u^2} \, dx$$

- **Program code:**

```
Int[v_*ArcTanh[u_],x_Symbol] :=
  Module[{w=Block[{ShowSteps=False,StepCounter=Null}, Int[v,x]]},
  w*ArcTanh[u] -
  Int[Regularize[w*D[u,x]/(1-u^2),x],x] /;
 InverseFunctionFreeQ[w,x]] /;
InverseFunctionFreeQ[u,x] &&
    Not[MatchQ[v, x^m_. /; FreeQ[m,x]]] &&
    FalseQ[FunctionOfLinear[v*ArcTanh[u],x]]
```