Integration Rules for

$$\int \left(\sin^j(z)\right)^m \left(A + B\sin^k(z)\right)\left(a + b\sin^k(z)\right)^n \, dz \text{ when } j^2 = 1 \bigwedge k^2 = 1 \bigwedge a^2 = b^2$$

Domain Map



Legend:

- The rule number in a colored region indicates the rule to use for integrals in that region.
- The rule number next to a colored line indicates the rule to use for integrals on that line.
- A white region or line indicates there is no rule for integrals in that region or on that line.
- A solid black line indicates integrals on that line are handled by rules in another section.
- A dashed black line on the border of a region indicates integrals on that border are handled by the rule for that region.
- The arrow(s) following a rule number indicates the direction the rule drives integrands in the n×m exponent plane.
- A ⋄ following a rule number indicates the rule transforms the integrand into a form handled by another section.
- A red (stop) disk indicates the terminal rule to use for the point at the center of the disk.
- A cyan disk indicates the non-terminal rule to use for the point at the center of the disk.

Integration Rules for $\int(A+B\sin^k(z))(a+b\sin^k(z))^n\,dz$ when $k^2=1 \wedge a^2=b^2$

Rule a: $\int \frac{A+B\,\text{Csc}[c+d\,x]}{a+b\,\text{Csc}[c+d\,x]}\,dx$

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{A+B z}{a+b z}=\frac{A}{a}-\frac{(b A-a B) z}{a (a+b z)}$
- **Note: The rule for integrands of the same form when $a^2-b^2\neq 0$ could subsume this rule, but the resulting antiderivative will look less like the integrand involving sines instead of cosecants.**
- **Rule a: If $a^2-b^2=0 \wedge b A-a B\neq 0$, then**

$$\int \frac{A+B\,\text{Csc}[c+d\,x]}{a+b\,\text{Csc}[c+d\,x]}\,dx \longrightarrow \frac{A\,x}{a}-\frac{b A-a B}{a}\int \frac{\text{Csc}[c+d\,x]}{a+b\,\text{Csc}[c+d\,x]}\,dx$$

- **Program code:**

```
Int[(A_.+B_.*sin[c_.+d_.*x_]^(-1))/(a_+b_.*sin[c_.+d_.*x_]^(-1)),x_Symbol] :=
  A*x/a - Dist[(b*A-a*B)/a,Int[sin[c+d*x]^(-1)/(a+b*sin[c+d*x]^(-1)),x]] /;
FreeQ[{a,b,c,d,A,B},x] && ZeroQ[a^2-b^2] && NonzeroQ[b*A-a*B]
```

Rules 13 – 14: $\int (A + B\,\mathrm{Csc}[c + d\,x])\,(a + b\,\mathrm{Csc}[c + d\,x])^n\,dx$

- **Derivation: Rule 6 with $m = 0$ and $k = -1$**
- **Rule 13: If $a^2 - b^2 = 0 \wedge b\,A - a\,B \neq 0 \wedge n < -1$, then**

$$\int (A + B\,\mathrm{Csc}[c + d\,x])\,(a + b\,\mathrm{Csc}[c + d\,x])^n\,dx \longrightarrow -\frac{(b\,A - a\,B)\,\mathrm{Cot}[c + d\,x]\,(a + b\,\mathrm{Csc}[c + d\,x])^n}{b\,d\,(2\,n + 1)} + \frac{1}{b^2\,(2\,n + 1)} \int (a\,A\,(2\,n + 1) - (b\,A - a\,B)\,(n + 1)\,\mathrm{Csc}[c + d\,x])\,(a + b\,\mathrm{Csc}[c + d\,x])^{n+1}\,dx$$

- **Program code:**

```
Int[(A_.+B_.*sin[c_.+d_.*x_]^(-1))*(a_+b_.*sin[c_.+d_.*x_]^(-1))^n_,x_Symbol] :=
  -(b*A-a*B)*Cot[c+d*x]*(a+b*Csc[c+d*x])^n/(b*d*(2*n+1)) +
  Dist[1/(b^2*(2*n+1)),
    Int[Sim[a*A*(2*n+1)-((b*A-a*B)*(n+1))*sin[c+d*x]^(-1),x]*(a+b*sin[c+d*x]^(-1))^(n+1),x]] /;
FreeQ[{a,b,c,d,A,B},x] && ZeroQ[a^2-b^2] && NonzeroQ[b*A-a*B] && RationalQ[n] && n<-1
```

- **Derivation: Rule 3 with $m = 0$ and $k = -1$**
- **Rule 14: If $a^2 - b^2 = 0 \wedge b\,A - a\,B \neq 0 \wedge n > 0$, then**

$$\int (A + B\,\mathrm{Csc}[c + d\,x])\,(a + b\,\mathrm{Csc}[c + d\,x])^n\,dx \longrightarrow -\frac{b\,B\,\mathrm{Cot}[c + d\,x]\,(a + b\,\mathrm{Csc}[c + d\,x])^{n-1}}{d\,n} + \frac{1}{n} \int (a\,A\,n + (a\,B\,(2\,n - 1) + b\,A\,n)\,\mathrm{Csc}[c + d\,x])\,(a + b\,\mathrm{Csc}[c + d\,x])^{n-1}\,dx$$

- **Program code:**

```
Int[(A_+B_.*sin[c_.+d_.*x_]^(-1))*(a_+b_.*sin[c_.+d_.*x_]^(-1))^n_,x_Symbol] :=
  -b*B*Cot[c+d*x]*(a+b*Csc[c+d*x])^(n-1)/(d*n) +
  Dist[1/n,
    Int[Sim[a*A*n+(a*B*(2*n-1)+b*A*n)*sin[c+d*x]^(-1),x]*(a+b*sin[c+d*x]^(-1))^(n-1),x]] /;
FreeQ[{a,b,c,d,A,B},x] && ZeroQ[a^2-b^2] && NonzeroQ[b*A-a*B] && RationalQ[n] && n>0
```

Integration Rules for

$$\int\left(\sin^j(z)\right)^m\left(A+B\sin^k(z)\right)\left(a+b\sin^k(z)\right)^n dz \text{ when } j^2=1\bigwedge k^2=1\bigwedge a^2=b^2$$

Rule c: $\int \frac{A+B\,\text{Sin}[c+d\,x]^k}{\text{Sin}[c+d\,x]^{\frac{k+1}{2}}\sqrt{a+b\,\text{Sin}[c+d\,x]^k}}\,dx$

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{A+B\,z^k}{z^{\frac{k+1}{2}}\sqrt{a+b\,z^k}} = \frac{A\sqrt{a+b\,z^k}}{a\,z^{\frac{k+1}{2}}} - \frac{(b\,A-a\,B)\,z^{\frac{k-1}{2}}}{a\sqrt{a+b\,z^k}}$
- **Rule c: If** $k^2=1\bigwedge a^2-b^2=0\bigwedge b\,A-a\,B\neq 0$**, then**

$$\int \frac{A+B\,\text{Sin}[c+d\,x]^k}{\text{Sin}[c+d\,x]^{\frac{k+1}{2}}\sqrt{a+b\,\text{Sin}[c+d\,x]^k}}\,dx \longrightarrow \frac{A}{a}\int\frac{\sqrt{a+b\,\text{Sin}[c+d\,x]^k}}{\text{Sin}[c+d\,x]^{\frac{k+1}{2}}}\,dx - \frac{b\,A-a\,B}{a}\int\frac{\text{Sin}[c+d\,x]^{\frac{k-1}{2}}}{\sqrt{a+b\,\text{Sin}[c+d\,x]^k}}\,dx$$

- **Program code:**

```
Int[(A_+B_.*sin[c_.+d_.*x_])/(sin[c_.+d_.*x_]*Sqrt[a_+b_.*sin[c_.+d_.*x_]]),x_Symbol] :=
  Dist[A/a,Int[Sqrt[a+b*sin[c+d*x]]/sin[c+d*x],x]] -
  Dist[(a*A-b*B)/b,Int[1/Sqrt[a+b*sin[c+d*x]],x]] /;
FreeQ[{a,b,c,d,A,B},x] && ZeroQ[a^2-b^2] && NonzeroQ[b*A-a*B]
```

```
Int[(A_+B_.*sin[c_.+d_.*x_]^(-1))/Sqrt[a_+b_.*sin[c_.+d_.*x_]^(-1)],x_Symbol] :=
  A/a*Int[Sqrt[a+b*sin[c+d*x]^(-1)],x] -
  (b*A-a*B)/a*Int[sin[c+d*x]^(-1)/Sqrt[a+b*sin[c+d*x]^(-1)],x] /;
FreeQ[{a,b,c,d,A,B},x] && ZeroQ[a^2-b^2] && NonzeroQ[b*A-a*B]
```

Rule d: $\int \frac{A + B \operatorname{Sin}[c + d x]}{\sqrt{\operatorname{Sin}[c + d x]} \sqrt{a + b \operatorname{Sin}[c + d x]}} \, dx$

- **Derivation: Algebraic expansion**
- **Rule d: If $a^2 - b^2 = 0 \land b A - a B \neq 0$, then**

$$\int \frac{A + B \operatorname{Sin}[c + d x]}{\sqrt{\operatorname{Sin}[c + d x]} \sqrt{a + b \operatorname{Sin}[c + d x]}} \, dx \to$$

$$\frac{B}{b} \int \frac{\sqrt{a + b \operatorname{Sin}[c + d x]}}{\sqrt{\operatorname{Sin}[c + d x]}} \, dx + \frac{b A - a B}{b} \int \frac{1}{\sqrt{\operatorname{Sin}[c + d x]} \sqrt{a + b \operatorname{Sin}[c + d x]}} \, dx$$

- **Program code:**

```
Int[(A_+B_.*sin[c_.+d_.*x_])/(Sqrt[sin[c_.+d_.*x_]]*Sqrt[a_+b_.*sin[c_.+d_.*x_]]),x_Symbol] :=
  Dist[B/b,Int[Sqrt[a+b*sin[c+d*x]]/Sqrt[sin[c+d*x]],x]] +
  Dist[(b*A-a*B)/b,Int[1/(Sqrt[sin[c+d*x]]*Sqrt[a+b*sin[c+d*x]]),x]] /;
FreeQ[{a,b,c,d,A,B},x] && ZeroQ[a^2-b^2] && NonzeroQ[b*A-a*B]
```

$$\int \left(\text{Sin}[c+d\,x]^j\right)^{m/2} (A+B\,\text{Csc}[c+d\,x]) (a+b\,\text{Csc}[c+d\,x])^{n/2}\,dx$$

- **Derivation: Rule 4 with** $j\,m=\frac{1}{2}$, $k=-1$ **and** $n=\frac{1}{2}$
- **Rule: If** $j^2=1 \land a^2-b^2=0 \land j\,m=\frac{1}{2}$, **then**

$$\int \left(\text{Sin}[c+d\,x]^j\right)^m (A+B\,\text{Csc}[c+d\,x]) \sqrt{a+b\,\text{Csc}[c+d\,x]}\,dx \longrightarrow$$

$$-\frac{2\,a\,A\,\text{Cos}[c+d\,x]}{d\left(\text{Sin}[c+d\,x]^j\right)^m \sqrt{a+b\,\text{Csc}[c+d\,x]}}+B\int \frac{\sqrt{a+b\,\text{Csc}[c+d\,x]}}{\left(\text{Sin}[c+d\,x]^j\right)^m}\,dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_*(A_+B_.*sin[c_.+d_.*x_]^(-1))*Sqrt[a_+b_.*sin[c_.+d_.*x_]^(-1)],x_Symbo
  -2*a*A*Cos[c+d*x]/(d*(Sin[c+d*x]^j)^m*Sqrt[a+b*Csc[c+d*x]]) +
  Dist[B,Int[Sqrt[a+b*sin[c+d*x]^(-1)]/(sin[c+d*x]^j)^m,x]] /;
FreeQ[{a,b,c,d,A,B},x] && OneQ[j^2] && ZeroQ[a^2-b^2] && RationalQ[m] && j*m==1/2
```

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{A+B\,z}{\sqrt{a+b\,z}}=\frac{B\sqrt{a+b\,z}}{b}+\frac{b\,A-a\,B}{b\sqrt{a+b\,z}}$
- **Rule: If** $j^2=1 \land a^2-b^2=0 \land j\,m=-\frac{1}{2} \bigwedge b\,A-a\,B\neq 0$, **then**

$$\int \frac{\left(\text{Sin}[c+d\,x]^j\right)^m (A+B\,\text{Csc}[c+d\,x])}{\sqrt{a+b\,\text{Csc}[c+d\,x]}}\,dx \longrightarrow$$

$$\frac{B}{b}\int \left(\text{sin}[c+d\,x]^j\right)^m \sqrt{a+b\,\text{Csc}[c+d\,x]}\,dx+\frac{b\,A-a\,B}{b}\int \frac{\left(\text{Sin}[c+d\,x]^j\right)^m}{\sqrt{a+b\,\text{Csc}[c+d\,x]}}\,dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_*(A_.+B_.*sin[c_.+d_.*x_]^(-1))/Sqrt[a_+b_.*sin[c_.+d_.*x_]^(-1)],x_Symb
  Dist[B/b,Int[(sin[c+d*x]^j)^m*Sqrt[a+b*sin[c+d*x]^(-1)],x]] +
  Dist[(b*A-a*B)/b,Int[(sin[c+d*x]^j)^m/Sqrt[a+b*sin[c+d*x]^(-1)],x]] /;
FreeQ[{a,b,c,d,A,B},x] && OneQ[j^2] && ZeroQ[a^2-b^2] && RationalQ[m] && j*m==-1/2 &&
  NonzeroQ[b*A-a*B]
```

- **Derivation:** Rule 5 with $j\,m = \frac{1}{2}$, $k = -1$ and $n = -\frac{1}{2}$

- **Rule:** If $j^2 = 1 \land a^2 - b^2 = 0 \land j\,m = \frac{1}{2} \bigwedge b\,A - a\,B \neq 0$, then

$$\int \frac{\left(\text{Sin}[c + d\,x]^j\right)^m\,(A + B\,\text{Csc}[c + d\,x])}{\sqrt{a + b\,\text{Csc}[c + d\,x]}}\,dx \longrightarrow$$

$$-\frac{2\,A\,\text{Cos}[c + d\,x]}{d\,\left(\text{Sin}[c + d\,x]^j\right)^m\,\sqrt{a + b\,\text{Csc}[c + d\,x]}} - \frac{b\,A - a\,B}{a}\int \frac{1}{\left(\text{Sin}[c + d\,x]^j\right)^m\,\sqrt{a + b\,\text{Csc}[c + d\,x]}}\,dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_*(A_+B_.*sin[c_.+d_.*x_]^(-1))/Sqrt[a_+b_.*sin[c_.+d_.*x_]^(-1)],x_Symbo
  -2*A*Cos[c+d*x]/(d*(Sin[c+d*x]^j)^m*Sqrt[a+b*Csc[c+d*x]]) -
  Dist[(b*A-a*B)/a,Int[1/((sin[c+d*x]^j)^m*Sqrt[a+b*sin[c+d*x]^(-1)]),x]] /;
FreeQ[{a,b,c,d,A,B},x] && OneQ[j^2] && ZeroQ[a^2-b^2] && RationalQ[m] && j*m==1/2 &&
  NonzeroQ[b*A-a*B]
```

Rules 9 – 10: $\int \left(\text{Sin}[c+d\,x]^j\right)^m \left(A+B\,\text{Sin}[c+d\,x]^k\right) \sqrt{a+b\,\text{Sin}[c+d\,x]^k}\,dx$

- **Derivation: Rule 4 with** $n = \frac{1}{2}$ **and** $a\,B\left(j\,k\,m+\frac{k+1}{2}\right)+b\,A\left(j\,k\,m+\frac{k+2}{2}\right)=0$
- **Derivation: Rule 3 with** $n = \frac{1}{2}$ **and** $a\,B\left(j\,k\,m+\frac{k+1}{2}\right)+b\,A\left(j\,k\,m+\frac{k+2}{2}\right)=0$
- **Rule 9a: If** $j^2=k^2=1 \land a^2-b^2=0 \bigwedge b\,A-a\,B\neq 0 \bigwedge j\,k\,m+\frac{k+1}{2}\neq 0 \bigwedge a\,B\left(j\,k\,m+\frac{k+1}{2}\right)+b\,A\left(j\,k\,m+\frac{k+2}{2}\right)=0$**, then**

$$\int \left(\text{Sin}[c+d\,x]^j\right)^m \left(A+B\,\text{Sin}[c+d\,x]^k\right) \sqrt{a+b\,\text{Sin}[c+d\,x]^k}\,dx \longrightarrow \frac{a\,A\,\text{Cos}[c+d\,x]\left(\text{Sin}[c+d\,x]^j\right)^{m+j\,k}}{d\left(j\,k\,m+\frac{k+1}{2}\right)\sqrt{a+b\,\text{Sin}[c+d\,x]^k}}$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_*(A_.+B_.*sin[c_.+d_.*x_]^k_.)*Sqrt[a_+b_.*sin[c_.+d_.*x_]^k_.],x_Symbol
  a*A*Cos[c+d*x]*(Sin[c+d*x]^j)^(m+j*k)/(d*(j*k*m+(k+1)/2)*Sqrt[a+b*Sin[c+d*x]^k]) /;
FreeQ[{a,b,c,d,A,B,m},x] && OneQ[j^2,k^2] && ZeroQ[a^2-b^2] && NonzeroQ[b*A-a*B] &&
  NonzeroQ[j*k*m+(k+1)/2] && ZeroQ[a*B*(j*k*m+(k+1)/2)+b*A*(j*k*m+(k+2)/2)]
```

- **Derivation: Rule 4 with** $n = \frac{1}{2}$
- **Rule 9b: If**
$j^2=k^2=1 \land a^2-b^2=0 \bigwedge b\,A-a\,B\neq 0 \bigwedge j\,k\,m+\frac{k+1}{2}\neq 0 \bigwedge j\,k\,m\leq -1 \bigwedge a\,B\left(j\,k\,m+\frac{k+1}{2}\right)+b\,A\left(j\,k\,m+\frac{k+2}{2}\right)\neq 0$**, then**

$$\int \left(\text{Sin}[c+d\,x]^j\right)^m \left(A+B\,\text{Sin}[c+d\,x]^k\right) \sqrt{a+b\,\text{Sin}[c+d\,x]^k}\,dx \longrightarrow \frac{a\,A\,\text{Cos}[c+d\,x]\left(\text{Sin}[c+d\,x]^j\right)^{m+j\,k}}{d\left(j\,k\,m+\frac{k+1}{2}\right)\sqrt{a+b\,\text{Sin}[c+d\,x]^k}}+$$
$$\frac{a\,B\left(j\,k\,m+\frac{k+1}{2}\right)+b\,A\left(j\,k\,m+\frac{k+2}{2}\right)}{a\left(j\,k\,m+\frac{k+1}{2}\right)}\int \left(\text{Sin}[c+d\,x]^j\right)^{m+j\,k}\sqrt{a+b\,\text{Sin}[c+d\,x]^k}\,dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(A_.+B_.*sin[c_.+d_.*x_]^k_.)*Sqrt[a_+b_.*sin[c_.+d_.*x_]^k_.],x_Symbo
  a*A*Cos[c+d*x]*(Sin[c+d*x]^j)^(m+j*k)/(d*(j*k*m+(k+1)/2)*Sqrt[a+b*Sin[c+d*x]^k]) +
  Dist[(a*B*(j*k*m+(k+1)/2)+b*A*(j*k*m+(k+2)/2))/(a*(j*k*m+(k+1)/2)),
    Int[(sin[c+d*x]^j)^(m+j*k)*Sqrt[a+b*sin[c+d*x]^k],x]] /;
FreeQ[{a,b,c,d,A,B},x] && OneQ[j^2,k^2] && ZeroQ[a^2-b^2] && NonzeroQ[b*A-a*B] &&
  RationalQ[m] && NonzeroQ[j*k*m+(k+1)/2] && j*k*m≤-1 &&
  NonzeroQ[a*B*(j*k*m+(k+1)/2)+b*A*(j*k*m+(k+2)/2)]
```

- **Derivation: Rule 3 with** $n = \frac{1}{2}$

- **Rule 10: If**
$j^2 = k^2 = 1 \wedge a^2 - b^2 = 0 \bigwedge b A - a B \neq 0 \bigwedge j k m + \frac{k+2}{2} \neq 0 \bigwedge j k m \geq -1 \bigwedge a B \left(j k m + \frac{k+1}{2}\right) + b A \left(j k m + \frac{k+2}{2}\right) \neq 0$, **then**

$$\int \left(\mathrm{Sin}[c+d\,x]^j\right)^m \left(A + B\,\mathrm{Sin}[c+d\,x]^k\right) \sqrt{a + b\,\mathrm{Sin}[c+d\,x]^k}\, dx \longrightarrow -\frac{b\,B\,\mathrm{Cos}[c+d\,x] \left(\mathrm{Sin}[c+d\,x]^j\right)^{m+j\,k}}{d \left(j\,k\,m + \frac{k+2}{2}\right) \sqrt{a + b\,\mathrm{Sin}[c+d\,x]^k}} +$$
$$\frac{a\,B \left(j\,k\,m + \frac{k+1}{2}\right) + b\,A \left(j\,k\,m + \frac{k+2}{2}\right)}{b \left(j\,k\,m + \frac{k+2}{2}\right)} \int \left(\mathrm{Sin}[c+d\,x]^j\right)^m \sqrt{a + b\,\mathrm{Sin}[c+d\,x]^k}\, dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_*(A_.+B_.*sin[c_.+d_.*x_]^k_.)*Sqrt[a_+b_.*sin[c_.+d_.*x_]^k_.],x_Symbol
  -b*B*Cos[c+d*x]*(Sin[c+d*x]^j)^(m+j*k)/(d*(j*k*m+(k+2)/2)*Sqrt[a+b*Sin[c+d*x]^k]) +
  Dist[(a*B*(j*k*m+(k+1)/2)+b*A*(j*k*m+(k+2)/2))/(b*(j*k*m+(k+2)/2)),
    Int[(sin[c+d*x]^j)^m*Sqrt[a+b*sin[c+d*x]^k],x]] /;
FreeQ[{a,b,c,d,A,B},x] && OneQ[j^2,k^2] && ZeroQ[a^2-b^2] && NonzeroQ[b*A-a*B] &&
  RationalQ[m] && NonzeroQ[j*k*m+(k+2)/2] && j*k*m>=-1 &&
  NonzeroQ[a*B*(j*k*m+(k+1)/2)+b*A*(j*k*m+(k+2)/2)]
```

Rules 11 – 12: $\int \frac{\left(\text{Sin}[c+d\,x]^j\right)^m \left(A+B\,\text{Sin}[c+d\,x]^k\right)}{\left(a+b\,\text{Sin}[c+d\,x]^k\right)^{j\,k\,m+\frac{k+3}{2}}}\,dx$

- **Derivation: Rule 5 with** $j\,k\,m+n+\frac{k+3}{2}=0$ **and** $a\,B\,(n+1)+b\,A\,n=0$
- **Rule 11a: If** $j^2=k^2=1\wedge a^2-b^2=0\bigwedge b\,A-a\,B\neq 0\bigwedge j\,k\,m+n+\frac{k+3}{2}=0\wedge n+1\neq 0\wedge a\,B\,(n+1)+b\,A\,n=0$, **then**

$$\int\left(\text{Sin}[c+d\,x]^j\right)^m\left(A+B\,\text{Sin}[c+d\,x]^k\right)\left(a+b\,\text{Sin}[c+d\,x]^k\right)^n dx\longrightarrow$$
$$-\frac{A\,\text{Cos}[c+d\,x]\left(\text{Sin}[c+d\,x]^j\right)^{m+j\,k}\left(a+b\,\text{Sin}[c+d\,x]^k\right)^n}{d\,(n+1)}$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_*(A_.+B_.*sin[c_.+d_.*x_]^k_.)*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol]
  -A*Cos[c+d*x]*(Sin[c+d*x]^j)^(m+j*k)*(a+b*Sin[c+d*x]^k)^n/(d*(n+1)) /;
FreeQ[{a,b,c,d,A,B,m,n},x] && OneQ[j^2,k^2] && ZeroQ[a^2-b^2] && NonzeroQ[b*A-a*B] &&
  ZeroQ[j*k*m+n+(k+3)/2] && NonzeroQ[n+1] && ZeroQ[a*B*(n+1)+b*A*n]
```

- **Derivation: Rule 5 with** $j\,k\,m+n+\frac{k+3}{2}=0$
- **Rule 11b: If** $j^2=k^2=1\wedge a^2-b^2=0\bigwedge b\,A-a\,B\neq 0\bigwedge j\,k\,m+n+\frac{k+3}{2}=0\wedge n>-1\wedge a\,B\,(n+1)+b\,A\,n\neq 0$, **then**

$$\int\left(\text{Sin}[c+d\,x]^j\right)^m\left(A+B\,\text{Sin}[c+d\,x]^k\right)\left(a+b\,\text{Sin}[c+d\,x]^k\right)^n dx\longrightarrow$$
$$-\frac{A\,\text{Cos}[c+d\,x]\left(\text{Sin}[c+d\,x]^j\right)^{m+j\,k}\left(a+b\,\text{Sin}[c+d\,x]^k\right)^n}{d\,(n+1)}+$$
$$\frac{a\,B\,(n+1)+b\,A\,n}{a\,(n+1)}\int\left(\text{Sin}[c+d\,x]^j\right)^{m+j\,k}\left(a+b\,\text{Sin}[c+d\,x]^k\right)^n dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_*(A_.+B_.*sin[c_.+d_.*x_]^k_.)*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol]
  -A*Cos[c+d*x]*(Sin[c+d*x]^j)^(m+j*k)*(a+b*Sin[c+d*x]^k)^n/(d*(n+1)) +
  Dist[(a*B*(n+1)+b*A*n)/(a*(n+1)),Int[(sin[c+d*x]^j)^(m+j*k)*(a+b*sin[c+d*x]^k)^n,x]] /;
FreeQ[{a,b,c,d,A,B},x] && OneQ[j^2,k^2] && ZeroQ[a^2-b^2] && NonzeroQ[b*A-a*B] &&
  RationalQ[m,n] && ZeroQ[j*k*m+n+(k+3)/2] && n>-1 && NonzeroQ[a*B*(n+1)+b*A*n]
```

- **Derivation: Rule 6 with** $j\,k\,m + n + \frac{k+3}{2} = 0$ **and** $b\,B\,(n+1) + a\,A\,n = 0$
- **Rule 12a: If** $j^2 = k^2 = 1 \land a^2 - b^2 = 0 \land b\,A - a\,B \neq 0 \land j\,k\,m + n + \frac{k+3}{2} = 0 \land 2\,n + 1 \neq 0 \land b\,B\,(n+1) + a\,A\,n = 0$**, then**

$$\int \left(\text{Sin}[c + d\,x]^j\right)^m \left(A + B\,\text{Sin}[c + d\,x]^k\right) \left(a + b\,\text{Sin}[c + d\,x]^k\right)^n dx \longrightarrow$$

$$-\frac{(b\,A - a\,B)\,\text{Cos}[c + d\,x] \left(\text{Sin}[c + d\,x]^j\right)^{m+j\,k} \left(a + b\,\text{Sin}[c + d\,x]^k\right)^n}{b\,d\,(2\,n + 1)}$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_*(A_.+B_.*sin[c_.+d_.*x_]^k_.)*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol]
  -(b*A-a*B)*Cos[c+d*x]*(Sin[c+d*x]^j)^(m+j*k)*(a+b*Sin[c+d*x]^k)^n/(b*d*(2*n+1)) /;
FreeQ[{a,b,c,d,A,B,m,n},x] && OneQ[j^2,k^2] && ZeroQ[a^2-b^2] && NonzeroQ[b*A-a*B] &&
  ZeroQ[j*k*m+n+(k+3)/2] && NonzeroQ[2*n+1] && ZeroQ[b*B*(n+1)+a*A*n]
```

- **Derivation: Rule 6 with** $j\,k\,m + n + \frac{k+3}{2} = 0$
- **Rule 12b: If** $j^2 = k^2 = 1 \land a^2 - b^2 = 0 \land b\,A - a\,B \neq 0 \land j\,k\,m + n + \frac{k+3}{2} = 0 \land n \leq -1 \land b\,B\,(n+1) + a\,A\,n \neq 0$**, then**

$$\int \left(\text{Sin}[c + d\,x]^j\right)^m \left(A + B\,\text{Sin}[c + d\,x]^k\right) \left(a + b\,\text{Sin}[c + d\,x]^k\right)^n dx \longrightarrow$$

$$-\frac{(b\,A - a\,B)\,\text{Cos}[c + d\,x] \left(\text{Sin}[c + d\,x]^j\right)^{m+j\,k} \left(a + b\,\text{Sin}[c + d\,x]^k\right)^n}{b\,d\,(2\,n + 1)} + \frac{b\,B\,(n+1) + a\,A\,n}{a^2\,(2\,n + 1)} \int \left(\text{Sin}[c + d\,x]^j\right)^m \left(a + b\,\text{Sin}[c + d\,x]^k\right)^{n+1} dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_*(A_.+B_.*sin[c_.+d_.*x_]^k_.)*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol]
  -(b*A-a*B)*Cos[c+d*x]*(Sin[c+d*x]^j)^(m+j*k)*(a+b*Sin[c+d*x]^k)^n/(b*d*(2*n+1)) +
  Dist[(b*B*(n+1)+a*A*n)/(a^2*(2*n+1)),Int[(sin[c+d*x]^j)^m*(a+b*sin[c+d*x]^k)^(n+1),x]] /;
FreeQ[{a,b,c,d,A,B},x] && OneQ[j^2,k^2] && ZeroQ[a^2-b^2] && NonzeroQ[b*A-a*B] &&
  RationalQ[m,n] && ZeroQ[j*k*m+n+(k+3)/2] && n≤-1 && NonzeroQ[b*B*(n+1)+a*A*n]
```

Rules 7 – 8: $\int \mathrm{Sin}[c+d\,x]^{\frac{k-1}{2}}\left(A+B\,\mathrm{Sin}[c+d\,x]^k\right)\left(a+b\,\mathrm{Sin}[c+d\,x]^k\right)^n\,dx$

- **Derivation: Rule 2 with** $j\,m=\frac{k-1}{2}$ **and** $a\,B\,n+b\,A\,(n+1)=0$
- **Rule: If** $k^2=1\wedge a^2-b^2=0\wedge a\,B\,n+b\,A\,(n+1)=0$**, then**

$$\int \mathrm{Sin}[c+d\,x]^{\frac{k-1}{2}}\left(A+B\,\mathrm{Sin}[c+d\,x]^k\right)\left(a+b\,\mathrm{Sin}[c+d\,x]^k\right)^n\,dx\longrightarrow$$

$$-\frac{B\,\mathrm{Cos}[c+d\,x]\,\mathrm{Sin}[c+d\,x]^{\frac{k-1}{2}}\left(a+b\,\mathrm{Sin}[c+d\,x]^k\right)^n}{d\,(n+1)}$$

- **Program code:**

```
Int[(A_.+B_.*sin[c_.+d_.*x_])*(a_+b_.*sin[c_.+d_.*x_])^n_.,x_Symbol] :=
  -B*Cos[c+d*x]*(a+b*Sin[c+d*x])^n/(d*(n+1)) /;
FreeQ[{a,b,c,d,A,B,n},x] && ZeroQ[a^2-b^2] && ZeroQ[a*B*n+b*A*(n+1)]
```

```
Int[sin[c_.+d_.*x_]^(-1)*(A_+B_.*sin[c_.+d_.*x_]^(-1))*(a_+b_.*sin[c_.+d_.*x_]^(-1))^n_.,x_Symbol] :
  -B*Cot[c+d*x]*(a+b*Csc[c+d*x])^n/(d*(n+1)) /;
FreeQ[{a,b,c,d,A,B,n},x] && ZeroQ[a^2-b^2] && ZeroQ[a*B*n+b*A*(n+1)]
```

- **Derivation: Rule 1 with** $j\,m=\frac{k-1}{2}$
- **Rule 7: If** $k^2=1\wedge a^2-b^2=0\wedge b\,A-a\,B\neq 0\wedge n\leq -1\wedge a\,B\,n+b\,A\,(n+1)\neq 0$**, then**

$$\int \mathrm{Sin}[c+d\,x]^{\frac{k-1}{2}}\left(A+B\,\mathrm{Sin}[c+d\,x]^k\right)\left(a+b\,\mathrm{Sin}[c+d\,x]^k\right)^n\,dx\longrightarrow$$

$$\frac{(b\,A-a\,B)\,\mathrm{Cos}[c+d\,x]\,\mathrm{Sin}[c+d\,x]^{\frac{k-1}{2}}\left(a+b\,\mathrm{Sin}[c+d\,x]^k\right)^n}{a\,d\,(2\,n+1)}+$$
$$\frac{a\,B\,n+b\,A\,(n+1)}{a\,b\,(2\,n+1)}\int \mathrm{Sin}[c+d\,x]^{\frac{k-1}{2}}\left(a+b\,\mathrm{Sin}[c+d\,x]^k\right)^{n+1}dx$$

- **Program code:**

```
Int[(A_.+B_.*sin[c_.+d_.*x_])*(a_+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  (b*A-a*B)*Cos[c+d*x]*(a+b*Sin[c+d*x])^n/(a*d*(2*n+1)) +
  Dist[(a*B*n+b*A*(n+1))/(a*b*(2*n+1)),Int[(a+b*sin[c+d*x])^(n+1),x]] /;
FreeQ[{a,b,c,d,A,B},x] && ZeroQ[a^2-b^2] && NonzeroQ[b*A-a*B] && RationalQ[n] && n≤-1 &&
  NonzeroQ[a*B*n+b*A*(n+1)]
```

```
Int[sin[c_.+d_.*x_]^(-1)*(A_+B_.*sin[c_.+d_.*x_]^(-1))*(a_+b_.*sin[c_.+d_.*x_]^(-1))^n_,x_Symbol] :=
  (b*A-a*B)*Cot[c+d*x]*(a+b*Csc[c+d*x])^n/(a*d*(2*n+1)) +
  Dist[(a*B*n+b*A*(n+1))/(a*b*(2*n+1)),Int[sin[c+d*x]^(-1)*(a+b*sin[c+d*x]^(-1))^(n+1),x]] /;
FreeQ[{a,b,c,d,A,B},x] && ZeroQ[a^2-b^2] && NonzeroQ[b*A-a*B] && RationalQ[n] && n≤-1 &&
  NonzeroQ[a*B*n+b*A*(n+1)]
```

- **Derivation: Rule 2 with** $j\,m = \frac{k-1}{2}$

- **Rule 8: If** $k^2 = 1 \land a^2 - b^2 = 0 \land b A - a B \neq 0 \land n > -1 \land n \neq 1 \land a B n + b A (n+1) \neq 0$**, then**

$$\int \sin[c + d x]^{\frac{k-1}{2}} \left(A + B \sin[c + d x]^k\right) \left(a + b \sin[c + d x]^k\right)^n dx \longrightarrow$$

$$-\frac{B \cos[c + d x] \sin[c + d x]^{\frac{k-1}{2}} \left(a + b \sin[c + d x]^k\right)^n}{d (n+1)} + \frac{a B n + b A (n+1)}{b (n+1)} \int \sin[c + d x]^{\frac{k-1}{2}} \left(a + b \sin[c + d x]^k\right)^n dx$$

- **Program code:**

```
Int[(A_.+B_.*sin[c_.+d_.*x_])*(a_+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  -B*Cos[c+d*x]*(a+b*Sin[c+d*x])^n/(d*(n+1)) +
   Dist[(a*B*n+b*A*(n+1))/(b*(n+1)),Int[(a+b*sin[c+d*x])^n,x]] /;
FreeQ[{a,b,c,d,A,B},x] && ZeroQ[a^2-b^2] && NonzeroQ[b*A-a*B] && RationalQ[n] &&
  n>-1 && n≠1 && NonzeroQ[a*B*n+b*A*(n+1)]
```

```
Int[sin[c_.+d_.*x_]^(-1)*(A_+B_.*sin[c_.+d_.*x_]^(-1))*(a_+b_.*sin[c_.+d_.*x_]^(-1))^n_,x_Symbol] :=
  -B*Cot[c+d*x]*(a+b*Csc[c+d*x])^n/(d*(n+1)) +
   Dist[(a*B*n+b*A*(n+1))/(b*(n+1)),Int[sin[c+d*x]^(-1)*(a+b*sin[c+d*x]^(-1))^n,x]] /;
FreeQ[{a,b,c,d,A,B},x] && ZeroQ[a^2-b^2] && NonzeroQ[b*A-a*B] && RationalQ[n] &&
  n>-1 && n≠1 && NonzeroQ[a*B*n+b*A*(n+1)]
```

Rules 1 – 6: $\int \left(\text{Sin}[c+d\,x]^j\right)^m \left(A+B\,\text{Sin}[c+d\,x]^k\right) \left(a+b\,\text{Sin}[c+d\,x]^k\right)^n\,dx$

- **Derivation: Recurrence 7**
- **Rule 1: If** $j^2=k^2=1\ \wedge\ a^2-b^2=0\ \wedge\ b\,A-a\,B\neq 0\ \wedge\ j\,k\,m>0\ \wedge\ n\leq -1$**, then**

$$\int \left(\text{Sin}[c+d\,x]^j\right)^m \left(A+B\,\text{Sin}[c+d\,x]^k\right) \left(a+b\,\text{Sin}[c+d\,x]^k\right)^n\,dx \longrightarrow$$

$$\frac{(b\,A-a\,B)\,\text{Cos}[c+d\,x]\,\left(\text{Sin}[c+d\,x]^j\right)^m \left(a+b\,\text{Sin}[c+d\,x]^k\right)^n}{a\,d\,(2\,n+1)}+$$

$$\frac{1}{a^2\,(2\,n+1)}\int \left(\text{Sin}[c+d\,x]^j\right)^{m-j\,k}\cdot$$

$$\left(-(b\,A-a\,B)\left(j\,k\,m+\frac{k-1}{2}\right)+\left(b\,B\,n+a\,A\,(n+1)+(a\,A-b\,B)\left(j\,k\,m+\frac{k-1}{2}\right)\right)\text{Sin}[c+d\,x]^k\right)$$

$$\left(a+b\,\text{Sin}[c+d\,x]^k\right)^{n+1}\,dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(A_.+B_.*sin[c_.+d_.*x_]^k_.)*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol
  (b*A-a*B)*Cos[c+d*x]*(Sin[c+d*x]^j)^m*(a+b*Sin[c+d*x]^k)^n/(a*d*(2*n+1)) +
  Dist[1/(a^2*(2*n+1)),
    Int[(sin[c+d*x]^j)^(m-j*k)*
      Sim[-(b*A-a*B)*(j*k*m+(k-1)/2)+(b*B*n+a*A*(n+1)+(a*A-b*B)*(j*k*m+(k-1)/2))*sin[c+d*x]^k,x]*
      (a+b*sin[c+d*x]^k)^(n+1),x]] /;
FreeQ[{a,b,c,d,A,B},x] && OneQ[j^2,k^2] && ZeroQ[a^2-b^2] && NonzeroQ[b*A-a*B] &&
  RationalQ[m,n] && j*k*m>0 && n≤-1 && Not[j*k*m==1 && n==-1]
```

- **Derivation: Recurrence 8**

- **Rule 2: If** $j^2 = k^2 = 1 \wedge a^2 - b^2 = 0 \bigwedge b A - a B \neq 0 \bigwedge j k m + n + \frac{k+1}{2} \neq 0 \bigwedge j k m > 0 \bigwedge -1 < n < 0$, **then**

$$\int \left(\text{Sin}[c + d x]^j\right)^m \left(A + B \,\text{Sin}[c + d x]^k\right) \left(a + b \,\text{Sin}[c + d x]^k\right)^n \, dx \longrightarrow$$

$$-\frac{B \,\text{Cos}[c + d x] \left(\text{Sin}[c + d x]^j\right)^m \left(a + b \,\text{Sin}[c + d x]^k\right)^n}{d \left(j k m + n + \frac{k+1}{2}\right)} +$$

$$\frac{1}{a \left(j k m + n + \frac{k+1}{2}\right)} \int \left(\text{Sin}[c + d x]^j\right)^{m-j k} \cdot$$

$$\left(a B \left(j k m + \frac{k-1}{2}\right) + \left(b B n + a A \left(j k m + n + \frac{k+1}{2}\right)\right) \text{Sin}[c + d x]^k\right) \left(a + b \,\text{Sin}[c + d x]^k\right)^n dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(A_.+B_.*sin[c_.+d_.*x_]^k_.)*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol
  -B*Cos[c+d*x]*(Sin[c+d*x]^j)^m*(a+b*Sin[c+d*x]^k)^n/(d*(j*k*m+n+(k+1)/2)) +
  Dist[1/(a*(j*k*m+n+(k+1)/2)),
    Int[(sin[c+d*x]^j)^(m-j*k)*
      Sim[a*B*(j*k*m+(k-1)/2)+(b*B*n+a*A*(j*k*m+n+(k+1)/2))*sin[c+d*x]^k,x]*
      (a+b*sin[c+d*x]^k)^n,x]] /;
FreeQ[{a,b,c,d,A,B},x] && OneQ[j^2,k^2] && ZeroQ[a^2-b^2] && NonzeroQ[b*A-a*B] &&
  RationalQ[m,n] && NonzeroQ[j*k*m+n+(k+1)/2] && j*k*m>0 && -1<n<0 && Not[j*m==1 && k==1]
```

- **Derivation: Recurrence 9**
- **Note: In the case $n = \frac{1}{2}$, this rule simplifies to rule 10.**
- **Rule 3: If $j^2 = k^2 = 1 \land a^2 - b^2 = 0 \bigwedge b A - a B \neq 0 \bigwedge j k m + n + \frac{k+1}{2} \neq 0 \bigwedge j k m \geq -1 \bigwedge n > 0 \bigwedge n \neq \frac{1}{2}$, then**

$$\int \left(\text{Sin}[c + d x]^j\right)^m \left(A + B \,\text{Sin}[c + d x]^k\right) \left(a + b \,\text{Sin}[c + d x]^k\right)^n dx \longrightarrow$$

$$-\frac{b B \,\text{Cos}[c + d x] \left(\text{Sin}[c + d x]^j\right)^{m+j k} \left(a + b \,\text{Sin}[c + d x]^k\right)^{n-1}}{d \left(j k m + n + \frac{k+1}{2}\right)} +$$

$$\frac{1}{j k m + n + \frac{k+1}{2}} \int \left(\text{Sin}[c + d x]^j\right)^m \cdot$$

$$\left(a A n + (a A + b B) \left(j k m + \frac{k+1}{2}\right) + \left(b A + a B n + (b A + a B) \left(j k m + n + \frac{k-1}{2}\right)\right) \text{Sin}[c + d x]^k\right)$$

$$\left(a + b \,\text{Sin}[c + d x]^k\right)^{n-1} dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(A_.+B_.*sin[c_.+d_.*x_]^k_.)*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_.,x_Symbo
  -b*B*Cos[c+d*x]*(Sin[c+d*x]^j)^(m+j*k)*(a+b*Sin[c+d*x]^k)^(n-1)/(d*(j*k*m+n+(k+1)/2)) +
  Dist[1/(j*k*m+n+(k+1)/2),
    Int[(sin[c+d*x]^j)^m*
      Sim[a*A*n+(a*A+b*B)*(j*k*m+(k+1)/2)+(b*A+a*B*n+(b*A+a*B)*(j*k*m+n+(k-1)/2))*sin[c+d*x]^k,x]*
      (a+b*sin[c+d*x]^k)^(n-1),x]] /;
FreeQ[{a,b,c,d,A,B},x] && OneQ[j^2,k^2] && ZeroQ[a^2-b^2] && NonzeroQ[b*A-a*B] &&
  RationalQ[m,n] && NonzeroQ[j*k*m+n+(k+1)/2] && j*k*m≥-1 && n>0 && n≠1/2 && Not[j*m==1 && k==1]
```

- **Derivation: Recurrence 10**
- **Note: In the case $n = \frac{1}{2}$, this rule simplifies to rule 9b.**
- **Rule 4: If $j^2 = k^2 = 1 \land a^2 - b^2 = 0 \bigwedge b A - a B \neq 0 \bigwedge j k m < -1 \bigwedge n > 0 \bigwedge n \neq \frac{1}{2}$, then**

$$\int \left(\text{Sin}[c + d x]^j\right)^m \left(A + B \,\text{Sin}[c + d x]^k\right) \left(a + b \,\text{Sin}[c + d x]^k\right)^n dx \longrightarrow$$

$$\frac{a A \,\text{Cos}[c + d x] \left(\text{Sin}[c + d x]^j\right)^{m+j k} \left(a + b \,\text{Sin}[c + d x]^k\right)^{n-1}}{d \left(j k m + \frac{k+1}{2}\right)} +$$

$$\frac{1}{j k m + \frac{k+1}{2}} \int \left(\text{Sin}[c + d x]^j\right)^{m+j k} \cdot$$

$$\left((b A + a B) \left(j k m + \frac{k+1}{2}\right) - b A (n-1) + \left(a A n + (a A + b B) \left(j k m + \frac{k+1}{2}\right)\right) \text{Sin}[c + d x]^k\right)$$

$$\left(a + b \,\text{Sin}[c + d x]^k\right)^{n-1} dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(A_.+B_.*sin[c_.+d_.*x_]^k_.)*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol
  a*A*Cos[c+d*x]*(Sin[c+d*x]^j)^(m+j*k)*(a+b*Sin[c+d*x]^k)^(n-1)/(d*(j*k*m+(k+1)/2)) +
  Dist[1/(j*k*m+(k+1)/2),
    Int[(sin[c+d*x]^j)^(m+j*k)*
      Sim[(b*A+a*B)*(j*k*m+(k+1)/2)-b*A*(n-1)+(a*A*n+(a*A+b*B)*(j*k*m+(k+1)/2))*sin[c+d*x]^k,x]*
      (a+b*sin[c+d*x]^k)^(n-1),x]] /;
FreeQ[{a,b,c,d,A,B},x] && OneQ[j^2,k^2] && ZeroQ[a^2-b^2] && NonzeroQ[b*A-a*B] &&
  RationalQ[m,n] && j*k*m<-1 && n>0 && n≠1/2
```

- **Derivation: Recurrence 11**

- **Rule 5: If** $j^2 = k^2 = 1 \land a^2 - b^2 = 0 \bigwedge b A - a B \neq 0 \bigwedge j k m + \frac{k+1}{2} \neq 0 \bigwedge j k m \leq -1 \bigwedge -1 < n < 0$**, then**

$$\int \left(\text{Sin}[c + d x]^j\right)^m \left(A + B \,\text{Sin}[c + d x]^k\right) \left(a + b \,\text{Sin}[c + d x]^k\right)^n \, dx \longrightarrow$$

$$\frac{A \,\text{Cos}[c + d x] \left(\text{Sin}[c + d x]^j\right)^{m+j k} \left(a + b \,\text{Sin}[c + d x]^k\right)^n}{d \left(j k m + \frac{k+1}{2}\right)} +$$

$$\frac{1}{a \left(j k m + \frac{k+1}{2}\right)} \int \left(\text{Sin}[c + d x]^j\right)^{m+j k} \cdot$$

$$\left(a B \left(j k m + \frac{k + 1}{2}\right) - b A n + a A \left(j k m + n + \frac{k + 3}{2}\right) \text{Sin}[c + d x]^k\right) \left(a + b \,\text{Sin}[c + d x]^k\right)^n \, dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(A_.+B_.*sin[c_.+d_.*x_]^k_.)*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_.,x_Symbo
  A*Cos[c+d*x]*(Sin[c+d*x]^j)^(m+j*k)*(a+b*Sin[c+d*x]^k)^n/(d*(j*k*m+(k+1)/2)) +
  Dist[1/(a*(j*k*m+(k+1)/2)),
    Int[(sin[c+d*x]^j)^(m+j*k)*
      Sim[a*B*(j*k*m+(k+1)/2)-b*A*n+a*A*(j*k*m+n+(k+3)/2)*sin[c+d*x]^k,x]*
      (a+b*sin[c+d*x]^k)^n,x]] /;
FreeQ[{a,b,c,d,A,B},x] && OneQ[j^2,k^2] && ZeroQ[a^2-b^2] && NonzeroQ[b*A-a*B] &&
  RationalQ[m,n] && NonzeroQ[j*k*m+(k+1)/2] && j*k*m≤-1 && -1<n<0
```

- **Derivation: Recurrence 12**

- **Rule 6: If** $j^2 = k^2 = 1 \bigwedge a^2 - b^2 \neq 0 \bigwedge b A - a B \neq 0 \bigwedge j k m < 0 \bigwedge n \leq -1$**, then**

$$\int \left(\text{Sin}[c + d x]^j\right)^m \left(A + B \,\text{Sin}[c + d x]^k\right) \left(a + b \,\text{Sin}[c + d x]^k\right)^n dx \longrightarrow$$

$$-\frac{(b A - a B) \,\text{Cos}[c + d x] \left(\text{Sin}[c + d x]^j\right)^{m+j k} \left(a + b \,\text{Sin}[c + d x]^k\right)^n}{b d (2 n + 1)} +$$

$$\frac{1}{b^2 (2 n + 1)} \int \left(\text{Sin}[c + d x]^j\right)^m \cdot$$

$$\left(a A (2 n + 1) + (a A - b B) \left(j k m + \frac{k + 1}{2}\right) - (b A - a B) \left(j k m + n + \frac{k + 3}{2}\right) \text{Sin}[c + d x]^k\right)$$

$$\left(a + b \,\text{Sin}[c + d x]^k\right)^{n+1} dx$$

- **Program code:**

```
Int[(sin[c_.+d_.*x_]^j_.)^m_.*(A_.+B_.*sin[c_.+d_.*x_]^k_.)*(a_+b_.*sin[c_.+d_.*x_]^k_.)^n_,x_Symbol
  -(b*A-a*B)*Cos[c+d*x]*(Sin[c+d*x]^j)^(m+j*k)*(a+b*Sin[c+d*x]^k)^n/(b*d*(2*n+1)) +
  Dist[1/(b^2*(2*n+1)),
    Int[(sin[c+d*x]^j)^m*
      Sim[a*A*(2*n+1)+(a*A-b*B)*(j*k*m+(k+1)/2)-(b*A-a*B)*(j*k*m+n+(k+3)/2)*sin[c+d*x]^k,x]*
      (a+b*sin[c+d*x]^k)^(n+1),x]] /;
FreeQ[{a,b,c,d,A,B},x] && OneQ[j^2,k^2] && ZeroQ[a^2-b^2] && NonzeroQ[b*A-a*B] &&
  RationalQ[m,n] && j*k*m<0 && n≤-1
```