$$\int (c\,(a+b\,x)^n)^m\,dx$$

- **Derivation: Reciprocal rule for integration**
- **Basis:** $\partial_x \frac{(a\,f[x]^n)^m}{f[x]^{m\,n}} = 0$
- **Rule: If** $m\,n+1=0$**, then**

$$\int (c\,(a+b\,x)^n)^m\,dx \longrightarrow \frac{(a+b\,x)\,(c\,(a+b\,x)^n)^m\,\mathrm{Log}[a+b\,x]}{b}$$

- **Program code:**

```
Int[(c_.*(a_.+b_.*x_)^n_.)^m_,x_Symbol] :=
  (a+b*x)*(c*(a+b*x)^n)^m*Log[a+b*x]/b /;
FreeQ[{a,b,c,m,n},x] && ZeroQ[m*n+1]
```

- **Derivation: Power rule for integration**
- **Basis:** $\partial_x \frac{(a\,f[x]^n)^m}{f[x]^{m\,n}} = 0$
- **Rule: If** $m\,n+1\neq 0$**, then**

$$\int (c\,(a+b\,x)^n)^m\,dx \longrightarrow \frac{(a+b\,x)\,(c\,(a+b\,x)^n)^m}{b\,(m\,n+1)}$$

- **Program code:**

```
Int[(c_.*(a_.+b_.*x_)^n_)^m_,x_Symbol] :=
  (a+b*x)*(c*(a+b*x)^n)^m/(b*(m*n+1)) /;
FreeQ[{a,b,c,m,n},x] && NonzeroQ[m*n+1]
```

$$\int (a\,x^m + b\,x^n)^p\,dx$$

- **Derivation: Algebraic simplification**
- **Basis:** $a\,z^m + b\,z^n = z^m\,(a + b\,z^{n-m})$
- **Note: Since** $m\,p + 1 = n - m$**, rule for** $\int x^{n-1}\,(a + b\,x^n)^p\,dx$ **applies.**
- **Rule: If** $p \in \mathbb{Z}\,\land\,m\,p + 1 = n - m$**, then**

$$\int (a\,x^m + b\,x^n)^p\,dx \longrightarrow \int x^{m\,p}\,(a + b\,x^{n-m})^p\,dx$$

- **Program code:**

```
Int[(a_.*x_^m_.+b_.*x_^n_.)^p_,x_Symbol] :=
  Int[x^(m*p)*(a+b*x^(n-m))^p,x] /;
FreeQ[{a,b,m,n},x] && IntegerQ[p] && ZeroQ[m*p+1-n+m] && Not[IntegersQ[m,n]]
```

- **Derivation: Algebraic simplification**
- **Basis:** $a\,z^m + b\,z^n = z^m\,(a + b\,z^{n-m})$
- **Note: Since** $m\,p + 1 = n - m$**, rule for** $\int \left(x^{\frac{n-1}{p}}\,(a + b\,x^n)\right)^p\,dx$ **applies.**
- **Rule: If** $p \in \mathbb{F}\,\land\,m\,p + 1 = n - m$**, then**

$$\int (a\,x^m + b\,x^n)^p\,dx \longrightarrow \int (x^m\,(a + b\,x^{n-m}))^p\,dx$$

- **Program code:**

```
Int[(a_.*x_^m_.+b_.*x_^n_.)^p_,x_Symbol] :=
  Int[(x^m*(a+b*x^(n-m)))^p,x] /;
FreeQ[{a,b,m,n},x] && FractionQ[p] && ZeroQ[m*p+1-n+m]
```

$$\int (a\,x^m + b\,x^n + c\,x^q)^p\,dx$$

- **Derivation: Algebraic simplification**
- **Basis:** $a\,z^m + b\,z^n + c\,z^q = z^m\,(a + b\,z^{n-m} + c\,z^{q-m})$
- **Rule: If** $m, n, q \in \mathbb{Z} \bigwedge p \in \mathbb{F} \bigwedge n \leq m \leq q$**, then**

$$\int (a\,x^m + b\,x^n + c\,x^q)^p\,dx \longrightarrow \int (x^m\,(a + b\,x^{n-m} + c\,x^{q-m}))^p\,dx$$

- **Program code:**

```
Int[(a_.*x_^m_.+b_.*x_^n_.+c_.*x_^q_.)^p_,x_Symbol] :=
  Int[(x^m*(a+b*x^(n-m)+c*x^(q-m)))^p,x] /;
FreeQ[{a,b,c},x] && IntegersQ[m,n,q] && FractionQ[p] && m<=n<=q
```

$$\int \frac{u\,x^m}{a\,x^n + b\,x^p}\,dx$$

- **Derivation: Algebraic simplification**
- **Basis:** $\frac{x^m}{a\,x^n+b\,x^p} = \frac{x^{m-n}}{a+b\,x^{p-n}}$
- **Rule: If** $m, n, p \in \mathbb{F} \wedge 0 < n \leq p$, **then**

$$\int \frac{u\,x^m}{a\,x^n + b\,x^p}\,dx \longrightarrow \int \frac{u\,x^{m-n}}{a + b\,x^{p-n}}\,dx$$

- **Program code:**

```
Int[u_.*x_^m_./(a_.*x_^n_.+b_.*x_^p_),x_Symbol] :=
  Int[u*x^(m-n)/(a+b*x^(p-n)),x] /;
FreeQ[{a,b},x] && FractionQ[{m,n,p}] && 0<n<=p
```

- **Derivation: Algebraic simplification**
- **Basis:** $\frac{x^m}{a\,x^n+b\,x^p} = \frac{x^{m-n}}{a+b\,x^{p-n}}$
- **Rule: If** $m, n, p \in \mathbb{F} \wedge 0 < n \leq p \wedge v$ is not a polynomial in x, **then**

$$\int \frac{u\,x^{m-n} + v}{a\,x^n + b\,x^p}\,dx \longrightarrow \int \frac{u\,x^{m-n}}{a + b\,x^{p-n}}\,dx + \int \frac{v}{a\,x^n + b\,x^p}\,dx$$

- **Program code:**

```
Int[(u_.*x_^m_.+v_)/(a_.*x_^n_.+b_.*x_^p_),x_Symbol] :=
  Int[u*x^(m-n)/(a+b*x^(p-n)),x] + Int[v/(a*x^n+b*x^p),x] /;
FreeQ[{a,b},x] && FractionQ[{m,n,p}] && 0<n<=p && Not[PolynomialQ[v,x]]
```

$$\int x^m\,(u+v+\cdots)\,dx$$

- **Derivation: Algebraic simplification**
- **Basis:** $x^m\,(u+v+\cdots) = x^m\,u + x^m\,v + \cdots$
- **Rule:**

$$\int x^m\,(u+v+\cdots)\,dx \longrightarrow \int (x^m\,u + x^m\,v + \cdots)\,dx$$

- **Program code:**

```
If[ShowSteps,
Int[x_^m_.*u_,x_Symbol] :=
  ShowStep["","Int[x^m*(u+v+⋯),x]","Int[x^m*u+x^m*v+⋯,x]",Hold[
  Int[Map[Function[x^m*#],u],x]]] /;
SimplifyFlag && FreeQ[m,x] && SumQ[u],

Int[x_^m_.*u_,x_Symbol] :=
  Int[Map[Function[x^m*#],u],x] /;
FreeQ[m,x] && SumQ[u]]
```

$$\int \frac{(a + b\,x^n)^p}{x}\,dx$$

- **Reference: CRC 276b**
- **Rule: If $a > 0$, then**

$$\int \frac{1}{x\sqrt{a + b\,x^n}}\,dx \longrightarrow -\frac{2}{n\sqrt{a}}\,\text{ArcTanh}\left[\frac{\sqrt{a + b\,x^n}}{\sqrt{a}}\right]$$

- **Program code:**

```
Int[1/(x_*Sqrt[a_+b_.*x_^n_.]),x_Symbol] :=
  -2*ArcTanh[Sqrt[a+b*x^n]/Rt[a,2]]/(n*Rt[a,2]) /;
FreeQ[{a,b,n},x] && PosQ[a]
```

- **Reference: CRC 277**
- **Rule: If $\neg\,(a > 0)$, then**

$$\int \frac{1}{x\sqrt{a + b\,x^n}}\,dx \longrightarrow \frac{2}{n\sqrt{-a}}\,\text{ArcTan}\left[\frac{\sqrt{a + b\,x^n}}{\sqrt{-a}}\right]$$

- **Program code:**

```
Int[1/(x_*Sqrt[a_+b_.*x_^n_.]),x_Symbol] :=
  2*ArcTan[Sqrt[a+b*x^n]/Rt[-a,2]]/(n*Rt[-a,2]) /;
FreeQ[{a,b,n},x] && NegQ[a]
```

- **Reference: G&R 2.110.1, CRC 88b**
- **Rule: If $p \in \mathbb{F} \wedge p > 0$, then**

$$\int \frac{(a + b\,x^n)^p}{x}\,dx \longrightarrow \frac{(a + b\,x^n)^p}{n\,p} + a\int \frac{(a + b\,x^n)^{p-1}}{x}\,dx$$

- **Program code:**

```
Int[(a_+b_.*x_^n_.)^p_/x_,x_Symbol] :=
  (a+b*x^n)^p/(n*p) +
  Dist[a,Int[(a+b*x^n)^(p-1)/x,x]] /;
FreeQ[{a,b,n},x] && FractionQ[p] && p>0
```

- **Reference: G&R 2.110.2, CRC 88d**

- **Rule: If $p \in \mathbb{F} \wedge p < -1$, then**

$$\int \frac{(a + b\, x^n)^p}{x}\, dx \longrightarrow -\frac{(a + b\, x^n)^{p+1}}{a\, n\, (p + 1)} + \frac{1}{a} \int \frac{(a + b\, x^n)^{p+1}}{x}\, dx$$

- **Program code:**

```
Int[(a_+b_.*x_^n_.)^p_/x_,x_Symbol] :=
  -(a+b*x^n)^(p+1)/(a*n*(p+1)) +
  Dist[1/a,Int[(a+b*x^n)^(p+1)/x,x]] /;
FreeQ[{a,b,n},x] && FractionQ[p] && p<-1
```

$$\int \frac{1}{(a+b x) \sqrt{c+d x}} \, dx$$

- **Reference: G&R 2.246.1', CRC 147a', A&S 3.3.30'**
- **Rule: If $\frac{b c-a d}{b} > 0$, then**

$$\int \frac{1}{(a+b x) \sqrt{c+d x}} \, dx \longrightarrow -\frac{2}{b \sqrt{\frac{b c-a d}{b}}} \text{ArcTanh}\left[\frac{\sqrt{c+d x}}{\sqrt{\frac{b c-a d}{b}}}\right]$$

- **Program code:**

```
Int[1/((a_.+b_.*x_)*Sqrt[c_.+d_.*x_]),x_Symbol] :=
  -2*ArcTanh[Sqrt[c+d*x]/Rt[(b*c-a*d)/b,2]]/(b*Rt[(b*c-a*d)/b,2]) /;
FreeQ[{a,b,c,d},x] && PosQ[(b*c-a*d)/b]
```

- **Reference: G&R 2.246.2, CRC 148, A&S 3.3.29**
- **Rule: If $\neg \left(\frac{b c-a d}{b} > 0\right)$, then**

$$\int \frac{1}{(a+b x) \sqrt{c+d x}} \, dx \longrightarrow \frac{2}{b \sqrt{\frac{a d-b c}{b}}} \text{ArcTan}\left[\frac{\sqrt{c+d x}}{\sqrt{\frac{a d-b c}{b}}}\right]$$

- **Program code:**

```
Int[1/((a_.+b_.*x_)*Sqrt[c_.+d_.*x_]),x_Symbol] :=
  2*ArcTan[Sqrt[c+d*x]/Rt[(a*d-b*c)/b,2]]/(b*Rt[(a*d-b*c)/b,2]) /;
FreeQ[{a,b,c,d},x] && NegQ[(b*c-a*d)/b]
```

$$\int \frac{1}{\sqrt{a+b\,x}\;\sqrt{c+d\,x}}\,dx$$

- **Rule: If $a+c=0 \wedge a>0$, then**

$$\int \frac{1}{\sqrt{a+b\,x}\;\sqrt{c+b\,x}}\,dx \longrightarrow \frac{1}{b}\,\text{ArcCosh}\left[\frac{b\,x}{a}\right]$$

- **Program code:**

```
Int[1/(Sqrt[a_+b_.*x_]*Sqrt[c_+b_.*x_]),x_Symbol] :=
  ArcCosh[b*x/a]/b /;
FreeQ[{a,b,c},x] && ZeroQ[a+c] && PositiveQ[a]
```

- **Rule: If $b+d=0 \wedge a+c>0$, then**

$$\int \frac{1}{\sqrt{a+b\,x}\;\sqrt{c+d\,x}}\,dx \longrightarrow \frac{1}{b}\,\text{ArcSin}\left[\frac{a-c+2\,b\,x}{a+c}\right]$$

- **Program code:**

```
Int[1/(Sqrt[a_+b_.*x_]*Sqrt[c_+d_.*x_]),x_Symbol] :=
  ArcSin[(a-c+2*b*x)/(a+c)]/b /;
FreeQ[{a,b,c,d},x] && ZeroQ[b+d] && PositiveQ[a+c]
```

- **Rule: If $a\,d-b\,c>0 \wedge d>0 \wedge b>0$, then**

$$\int \frac{1}{\sqrt{a+b\,x}\;\sqrt{c+d\,x}}\,dx \longrightarrow \frac{2}{\sqrt{b}\;\sqrt{d}}\,\text{ArcSinh}\left[\frac{\sqrt{b}\;\sqrt{c+d\,x}}{\sqrt{a\,d-b\,c}}\right]$$

- **Program code:**

```
Int[1/(Sqrt[a_.+b_.*x_]*Sqrt[c_.+d_.*x_]),x_Symbol] :=
  2/(Rt[b,2]*Sqrt[d])*ArcSinh[Rt[b,2]*Sqrt[c+d*x]/Sqrt[a*d-b*c]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a*d-b*c] && PositiveQ[d/(a*d-b*c)] && Not[NegativeQ[a*d-b*c]] && PosQ
```

- **Rule: If** $a\,d-b\,c>0\wedge d>0\wedge\neg(b>0)$**, then**

$$\int\frac{1}{\sqrt{a+b\,x}\,\sqrt{c+d\,x}}\,dx\;\longrightarrow\;\frac{2}{\sqrt{-b}\,\sqrt{d}}\,\text{ArcSin}\Big[\frac{\sqrt{-b}\,\sqrt{c+d\,x}}{\sqrt{a\,d-b\,c}}\Big]$$

- **Program code:**

```
Int[1/(Sqrt[a_.+b_.*x_]*Sqrt[c_.+d_.*x_]),x_Symbol] :=
  2/(Rt[-b,2]*Sqrt[d])*ArcSin[Rt[-b,2]*Sqrt[c+d*x]/Sqrt[a*d-b*c]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a*d-b*c] && PositiveQ[d/(a*d-b*c)] && Not[NegativeQ[a*d-b*c]] && NegQ
```

- **Rule: If** $a\,d-b\,c\neq 0\wedge b\,d>0$**, then**

$$\int\frac{1}{\sqrt{a+b\,x}\,\sqrt{c+d\,x}}\,dx\;\longrightarrow\;\frac{2}{\sqrt{b\,d}}\,\text{ArcTanh}\Big[\frac{\sqrt{b\,d}\,\sqrt{a+b\,x}}{b\,\sqrt{c+d\,x}}\Big]$$

- **Program code:**

```
Int[1/(Sqrt[a_.+b_.*x_]*Sqrt[c_.+d_.*x_]),x_Symbol] :=
  2/Rt[b*d,2]*ArcTanh[Rt[b*d,2]*Sqrt[a+b*x]/(b*Sqrt[c+d*x])] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a*d-b*c] && PosQ[b*d]
```

- **Rule: If** $a\,d-b\,c\neq 0\wedge\neg(b\,d>0)$**, then**

$$\int\frac{1}{\sqrt{a+b\,x}\,\sqrt{c+d\,x}}\,dx\;\longrightarrow\;\frac{2}{\sqrt{-b\,d}}\,\text{ArcTan}\Big[\frac{\sqrt{-b\,d}\,\sqrt{a+b\,x}}{b\,\sqrt{c+d\,x}}\Big]$$

- **Program code:**

```
Int[1/(Sqrt[a_.+b_.*x_]*Sqrt[c_.+d_.*x_]),x_Symbol] :=
  2/Rt[-b*d,2]*ArcTan[Rt[-b*d,2]*Sqrt[a+b*x]/(b*Sqrt[c+d*x])] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[a*d-b*c] && NegQ[b*d]
```

$$\int (a+b\,x)^m\,(c+d\,x)^n\,dx$$

- **Derivation: Algebraic simplification**
- **Basis: If** $a+c\geq 0$**, then** $(a+z)^m\,(c-z)^m = (a\,c-(a-c)\,z-z^2)^m$
- **Rule: If** $m\in\mathbb{F}\,\land\,b+d=0\,\land\,a+c>0$**, then**

$$\int (a+b\,x)^m\,(c+d\,x)^m\,dx \longrightarrow \int \left(a\,c+(a\,d+b\,c)\,x+b\,d\,x^2\right)^m\,dx$$

- **Program code:**

```
Int[(a_.+b_.*x_)^m_*(c_+d_.*x_)^m_,x_Symbol] :=
  Int[(a*c+(a*d+b*c)*x+b*d*x^2)^m,x] /;
FreeQ[{a,b,c,d},x] && FractionQ[m] && ZeroQ[b+d] && PositiveQ[a+c]
```

- **Derivation: Algebraic simplification**
- **Basis: If** $b\,c-a\,d=0$ **and** n **is an integer, then** $(a+b\,x)^m\,(c+d\,x)^n = \left(\frac{d}{b}\right)^n (a+b\,x)^{m+n}$
- **Rule: If** $b\,c-a\,d=0\,\land\,m\notin\mathbb{Z}\,\land\,n\in\mathbb{Z}$**, then**

$$\int (a+b\,x)^m\,(c+d\,x)^n\,dx \longrightarrow \left(\frac{d}{b}\right)^n \int (a+b\,x)^{m+n}\,dx$$

- **Program code:**

```
Int[(a_.+b_.*x_)^m_.*(c_.+d_.*x_)^n_.,x_Symbol] :=
  Dist[(d/b)^n,Int[(a+b*x)^(m+n),x]] /;
FreeQ[{a,b,c,d,m},x] && ZeroQ[b*c-a*d] && Not[IntegerQ[m]] && IntegerQ[n]
```

- **Rule: If** $b\,c-a\,d=0\,\land\,m\notin\mathbb{Z}\,\land\,n\notin\mathbb{Z}\,\land\,m+n+1=0$**, then**

$$\int (a+b\,x)^m\,(c+d\,x)^n\,dx \longrightarrow \frac{(a+b\,x)^{m+1}\,(c+d\,x)^n\,\mathrm{Log}[a+b\,x]}{b}$$

- **Program code:**

```
Int[(a_.+b_.*x_)^m_*(c_.+d_.*x_)^n_,x_Symbol] :=
  (a+b*x)^(m+1)*(c+d*x)^n*Log[a+b*x]/b /;
FreeQ[{a,b,c,d,m,n},x] && ZeroQ[b*c-a*d] && Not[IntegerQ[m]] && Not[IntegerQ[n]] && ZeroQ[m+n+1] &&
(LeafCount[a+b*x]<LeafCount[c+d*x] || LeafCount[a+b*x]==LeafCount[c+d*x] && PosQ[a])
```

- **Rule: If** $b c - a d = 0 \wedge m \notin \mathbb{Z} \wedge n \notin \mathbb{Z} \wedge m + n + 1 \neq 0$, **then**

$$\int (a + b x)^m (c + d x)^n \, dx \longrightarrow \frac{(a + b x)^{m+1} (c + d x)^n}{b (m + n + 1)}$$

- **Program code:**

```
Int[(a_.+b_.*x_)^m_*(c_.+d_.*x_)^n_,x_Symbol] :=
  (a+b*x)^(m+1)*(c+d*x)^n/(b*(m+n+1)) /;
FreeQ[{a,b,c,d,m,n},x] && ZeroQ[b*c-a*d] && Not[IntegerQ[m]] && Not[IntegerQ[n]] &&
NonzeroQ[m+n+1]
```

- **Reference: G&R 2.155, CRC 59a**

- **Rule: If** $b c - a d \neq 0 \wedge m + n + 2 = 0 \wedge n + 1 \neq 0$, **then**

$$\int (a + b x)^m (c + d x)^n \, dx \longrightarrow -\frac{(a + b x)^{m+1} (c + d x)^{n+1}}{(n + 1) (b c - a d)}$$

- **Program code:**

```
Int[(a_.+b_.*x_)^m_.*(c_.+d_.*x_.)^n_,x_Symbol] :=
  -(a+b*x)^(m+1)*(c+d*x)^(n+1)/((n+1)*(b*c-a*d)) /;
FreeQ[{a,b,c,d,m,n},x] && NonzeroQ[b*c-a*d] && ZeroQ[m+n+2] && NonzeroQ[n+1]
```

- **Reference: G&R 2.151, CRC 59b**

- **Rule: If** $b c - a d \neq 0 \wedge n \in \mathbb{F} \wedge n > 0$, **then**

$$\int \frac{(c + d x)^n}{a + b x} \, dx \longrightarrow \frac{(c + d x)^n}{b n} + \frac{b c - a d}{b} \int \frac{(c + d x)^{n-1}}{a + b x} \, dx$$

- **Program code:**

```
Int[(c_.+d_.*x_)^n_./(a_.+b_.*x_),x_Symbol] :=
  (c+d*x)^n/(b*n) +
  Dist[(b*c-a*d)/b,Int[(c+d*x)^(n-1)/(a+b*x),x]] /;
FreeQ[{a,b,c,d},x] && NonzeroQ[b*c-a*d] && FractionQ[n] && n>0
```

- **Reference: G&R 2.151, CRC 59b**
- **Note: Experimental!**
- **Rule: If** $b\,c-a\,d\neq 0\,\land\,m\notin\mathbb{Z}$**, then**

$$\int(a+b\,x)^m\,(c+d\,x)\,dx\,\longrightarrow\,\frac{(a+b\,x)^{m+1}\,(c+d\,x)}{b\,(m+2)}+\frac{b\,c-a\,d}{b\,(m+2)}\int(a+b\,x)^m\,dx$$

- **Program code:**

```
Int[(a_.+b_.*x_)^m_*(c_.+d_.*x_),x_Symbol] :=
  (a+b*x)^(m+1)*(c+d*x)/(b*(m+2)) +
  Dist[(b*c-a*d)/(b*(m+2)),Int[(a+b*x)^m,x]] /;
FreeQ[{a,b,c,d,m},x] && NonzeroQ[b*c-a*d] && Not[IntegerQ[m]]
```

- **Reference: G&R 2.155, CRC 59a**
- **Rule: If** $b\,c-a\,d\neq 0\,\land\,m+n+1\neq 0\,\land\,n>0\,\land\,m\notin\mathbb{Z}\,\land\,(n\in\mathbb{Z}\,\lor\,(m\in\mathbb{F}\,\land\,(n\leq m\,\lor\,-1\leq m<0)))$**, then**

$$\int(a+b\,x)^m\,(c+d\,x)^n\,dx\,\longrightarrow\,\frac{(a+b\,x)^{m+1}\,(c+d\,x)^n}{b\,(m+n+1)}+\frac{n\,(b\,c-a\,d)}{b\,(m+n+1)}\int(a+b\,x)^m\,(c+d\,x)^{n-1}\,dx$$

- **Program code:**

```
Int[(a_.+b_.*x_)^m_*(c_.+d_.*x_)^n_,x_Symbol] :=
  (a+b*x)^(m+1)*(c+d*x)^n/(b*(m+n+1)) +
  Dist[n*(b*c-a*d)/(b*(m+n+1)),Int[(a+b*x)^m*(c+d*x)^(n-1),x]] /;
FreeQ[{a,b,c,d,m},x] && NonzeroQ[b*c-a*d] && NonzeroQ[m+n+1] && RationalQ[n] && n>0 &&
Not[IntegerQ[m]] && (IntegerQ[n] || FractionQ[m] && (n<=m || -1<=m<0))
```

- **Reference: G&R 2.155, CRC 59a**
- **Rule: If** $b\,c-a\,d\neq 0\,\land\,n<-1\,\land\,\neg\,(m\in\mathbb{Z}\,\land\,n\in\mathbb{Z})\,\land\,(m\notin\mathbb{Q}\,\lor\,n\geq m\,\lor\,-1\leq m<0)$**, then**

$$\int(a+b\,x)^m\,(c+d\,x)^n\,dx\,\longrightarrow\,-\frac{(a+b\,x)^{m+1}\,(c+d\,x)^{n+1}}{(n+1)\,(b\,c-a\,d)}+\frac{b\,(m+n+2)}{(n+1)\,(b\,c-a\,d)}\int(a+b\,x)^m\,(c+d\,x)^{n+1}\,dx$$

- **Program code:**

```
Int[(a_.+b_.*x_)^m_.*(c_.+d_.*x_)^n_,x_Symbol] :=
  -(a+b*x)^(m+1)*(c+d*x)^(n+1)/((n+1)*(b*c-a*d)) +
  Dist[b*(m+n+2)/((n+1)*(b*c-a*d)),Int[(a+b*x)^m*(c+d*x)^(n+1),x]] /;
FreeQ[{a,b,c,d,m},x] && NonzeroQ[b*c-a*d] && RationalQ[n] && n<-1 && Not[IntegersQ[m,n]] &&
(Not[RationalQ[m]] || n>=m || -1<=m<0)
```

- **Reference: G&R 2.155, CRC 59a**

- **Rule: If** $b\,c-a\,d\neq 0\wedge n+1\neq 0\wedge n\notin\mathbb{Q}\wedge m+n<-1$**, then**

$$\int(a+b\,x)^m\,(c+d\,x)^n\,dx\;\longrightarrow\;-\frac{(a+b\,x)^{m+1}\,(c+d\,x)^{n+1}}{(n+1)\,(b\,c-a\,d)}+\frac{b\,(m+n+2)}{(n+1)\,(b\,c-a\,d)}\int(a+b\,x)^m\,(c+d\,x)^{n+1}\,dx$$

- **Program code:**

```
Int[(a_.+b_.*x_)^m_.*(c_.+d_.*x_)^n_,x_Symbol] :=
  -(a+b*x)^(m+1)*(c+d*x)^(n+1)/((n+1)*(b*c-a*d)) +
  Dist[b*(m+n+2)/((n+1)*(b*c-a*d)),Int[(a+b*x)^m*(c+d*x)^(n+1),x]] /;
FreeQ[{a,b,c,d,m,n},x] && NonzeroQ[b*c-a*d] && NonzeroQ[n+1] && Not[RationalQ[n]] &&
RationalQ[m+n] && Simplify[m+n]<-1
```

- **Reference: G&R 2.153.3, CRC 59c**

- **Rule: If** $n\notin\mathbb{Z}$**, then**

$$\int(a+b\,x)\,(c+d\,x)^n\,dx\;\longrightarrow\;\frac{(a+b\,x)\,(c+d\,x)^{n+1}}{d\,(n+1)}-\frac{b}{d\,(n+1)}\int(c+d\,x)^{n+1}\,dx$$

- **Program code:**

```
Int[(a_.+b_.*x_)*(c_.+d_.*x_)^n_,x_Symbol] :=
  (a+b*x)*(c+d*x)^(n+1)/(d*(n+1)) -
  Dist[b/(d*(n+1)),Int[(c+d*x)^(n+1),x]] /;
FreeQ[{a,b,c,d,n},x] && Not[IntegerQ[n]]
```

- **Reference: G&R 2.153.3, CRC 59c**

- **Rule: If** $\neg\,(m\in\mathbb{Z}\wedge n\in\mathbb{Z})\wedge m>0\wedge n<-1$**, then**

$$\int(a+b\,x)^m\,(c+d\,x)^n\,dx\;\longrightarrow\;\frac{(a+b\,x)^m\,(c+d\,x)^{n+1}}{d\,(n+1)}-\frac{b\,m}{d\,(n+1)}\int(a+b\,x)^{m-1}\,(c+d\,x)^{n+1}\,dx$$

- **Program code:**

```
Int[(a_.+b_.*x_)^m_*(c_.+d_.*x_)^n_,x_Symbol] :=
  (a+b*x)^m*(c+d*x)^(n+1)/(d*(n+1)) -
  Dist[b*m/(d*(n+1)),Int[(a+b*x)^(m-1)*(c+d*x)^(n+1),x]] /;
FreeQ[{a,b,c,d},x] && RationalQ[{m,n}] && Not[IntegersQ[m,n]] && m>0 && n<-1
```

- **Derivation: Integration by substitution**

- **Basis: If $n, p \in \mathbb{Z}$, then** $\frac{(a+b x)^{n/p}}{c+d x} = p \frac{\left((a+b x)^{1/p}\right)^{n+p-1}}{b c-a d+d \left((a+b x)^{1/p}\right)^{p}} \partial_x (a+b x)^{1/p}$

- **Rule: If $-1 < m < 0$, then**

$$\int \frac{(a+b x)^{n/p}}{c+d x} \, dx \longrightarrow p \operatorname{Subst}\left[\int \frac{x^{n+p-1}}{b c-a d+d x^p} \, dx, x, (a+b x)^{1/p}\right]$$

- **Program code:**

```
Int[(a_.+b_.*x_)^m_/(c_+d_.*x_),x_Symbol] :=
  Module[{p=Denominator[m]},
  Dist[p,Subst[Int[x^(m*p+p-1)/(b*c-a*d+d*x^p),x],x,(a+b*x)^(1/p)]]] /;
FreeQ[{a,b,c,d},x] && RationalQ[m] && -1<m<0
```

- **Derivation: Integration by substitution**

- **Basis: If $n, p \in \mathbb{Z}$, then** $\frac{(a+b x)^{n/p}}{c+d x} = p \frac{\left(\frac{(a+b x)^{1/p}}{(c+d x)^{1/p}}\right)^{m+p-1}}{b-d \left(\frac{(a+b x)^{1/p}}{(c+d x)^{1/p}}\right)^{p}} \partial_x \frac{(a+b x)^{1/p}}{(c+d x)^{1/p}}$

- **Rule: If $-1 < m < 0 \bigwedge m+n = -1$, then**

$$\int (a+b x)^{m/p} (c+d x)^n \, dx \longrightarrow p \operatorname{Subst}\left[\int \frac{x^{m+p-1}}{b-d x^p} \, dx, x, \frac{(a+b x)^{1/p}}{(c+d x)^{1/p}}\right]$$

- **Program code:**

```
Int[(a_.+b_.*x_)^m_*(c_.+d_.*x_)^n_,x_Symbol] :=
  Module[{p=Denominator[m]},
  Dist[p,Subst[Int[x^(m*p+p-1)/(b-d*x^p),x],x,(a+b*x)^(1/p)/(c+d*x)^(1/p)]]] /;
FreeQ[{a,b,c,d},x] && RationalQ[{m,n}] && -1<m<0 && m+n==-1
```

$$\int \frac{(a+b\,x)^n\,(c+d\,x)^n}{x}\,dx$$

- **Rule: If $a\,c>0$, then**

$$\int \frac{1}{x\,\sqrt{a+b\,x}\,\sqrt{c+d\,x}}\,dx \longrightarrow -\frac{2}{\sqrt{a\,c}}\,\text{ArcTanh}\Big[\frac{\sqrt{a\,c}\,\sqrt{a+b\,x}}{a\,\sqrt{c+d\,x}}\Big]$$

- **Program code:**

```
Int[1/(x_*Sqrt[a_+b_.*x_]*Sqrt[c_+d_.*x_]),x_Symbol] :=
  -2*ArcTanh[Rt[a*c,2]*Sqrt[a+b*x]/(a*Sqrt[c+d*x])]/Rt[a*c,2] /;
FreeQ[{a,b,c,d},x] && PosQ[a*c]
```

- **Rule: If $\neg\,(a\,c>0)$, then**

$$\int \frac{1}{x\,\sqrt{a+b\,x}\,\sqrt{c+d\,x}}\,dx \longrightarrow -\frac{2}{\sqrt{-a\,c}}\,\text{ArcTan}\Big[\frac{\sqrt{-a\,c}\,\sqrt{a+b\,x}}{a\,\sqrt{c+d\,x}}\Big]$$

- **Program code:**

```
Int[1/(x_*Sqrt[a_+b_.*x_]*Sqrt[c_+d_.*x_]),x_Symbol] :=
  -2*ArcTan[Rt[-a*c,2]*Sqrt[a+b*x]/(a*Sqrt[c+d*x])]/Rt[-a*c,2] /;
FreeQ[{a,b,c,d},x] && NegQ[a*c]
```

- **Reference: G&R 2.265b**
- **Rule: If $n\in\mathbb{F}\,\wedge\,n>0\,\wedge\,a\,d+b\,c=0$, then**

$$\int \frac{(a+b\,x)^n\,(c+d\,x)^n}{x}\,dx \longrightarrow \frac{(a+b\,x)^n\,(c+d\,x)^n}{2\,n}+a\,c\int \frac{(a+b\,x)^{n-1}\,(c+d\,x)^{n-1}}{x}\,dx$$

- **Program code:**

```
Int[(a_+b_.*x_)^n_*(c_+d_.*x_)^n_/x_,x_Symbol] :=
  (a+b*x)^n*(c+d*x)^n/(2*n) +
  Dist[a*c,Int[(a+b*x)^(n-1)*(c+d*x)^(n-1)/x,x]] /;
FreeQ[{a,b,c,d},x] && FractionQ[n] && n>0 && ZeroQ[a*d+b*c]
```

- **Rule: If $n \in \mathbb{F} \land n > 0 \land a d + b c \neq 0$, then**

$$\int \frac{(a+b x)^n (c+d x)^n}{x} \, dx \longrightarrow \frac{(a+b x)^n (c+d x)^n}{2 n} + \frac{(a d+b c)}{2} \int (a+b x)^{n-1} (c+d x)^{n-1} \, dx + a c \int \frac{(a+b x)^{n-1} (c+d x)^{n-1}}{x} \, dx$$

- **Program code:**

```
Int[(a_+b_.*x_)^n_*(c_+d_.*x_)^n_/x_,x_Symbol] :=
  (a+b*x)^n*(c+d*x)^n/(2*n) +
  Dist[(a*d+b*c)/2,Int[(a+b*x)^(n-1)*(c+d*x)^(n-1),x]] +
  Dist[a*c,Int[(a+b*x)^(n-1)*(c+d*x)^(n-1)/x,x]] /;
FreeQ[{a,b,c,d},x] && FractionQ[n] && n>0 && NonzeroQ[a*d+b*c]
```

- **Reference: G&R 2.268b, CRC 122**

- **Rule: If $n \in \mathbb{F} \land n < -1 \land a d + b c = 0$, then**

$$\int \frac{(a+b x)^n (c+d x)^n}{x} \, dx \longrightarrow -\frac{(a+b x)^{n+1} (c+d x)^{n+1}}{2 a c (n+1)} + \frac{1}{a c} \int \frac{(a+b x)^{n+1} (c+d x)^{n+1}}{x} \, dx$$

- **Program code:**

```
Int[(a_+b_.*x_)^n_*(c_+d_.*x_)^n_/x_,x_Symbol] :=
  -(a+b*x)^(n+1)*(c+d*x)^(n+1)/(2*a*c*(n+1)) +
  Dist[1/(a*c),Int[(a+b*x)^(n+1)*(c+d*x)^(n+1)/x,x]] /;
FreeQ[{a,b,c,d},x] && FractionQ[n] && n<-1 && ZeroQ[a*d+b*c]
```

- **Rule: If $n \in \mathbb{F} \land n < -1 \land a d + b c \neq 0$, then**

$$\int \frac{(a+b x)^n (c+d x)^n}{x} \, dx \longrightarrow -\frac{(a+b x)^{n+1} (c+d x)^{n+1}}{2 a c (n+1)} - \frac{a d+b c}{2 a c} \int (a+b x)^n (c+d x)^n \, dx + \frac{1}{a c} \int \frac{(a+b x)^{n+1} (c+d x)^{n+1}}{x} \, dx$$

- **Program code:**

```
Int[(a_+b_.*x_)^n_*(c_+d_.*x_)^n_/x_,x_Symbol] :=
  -(a+b*x)^(n+1)*(c+d*x)^(n+1)/(2*a*c*(n+1)) -
  Dist[(a*d+b*c)/(2*a*c),Int[(a+b*x)^n*(c+d*x)^n,x]] +
  Dist[1/(a*c),Int[(a+b*x)^(n+1)*(c+d*x)^(n+1)/x,x]] /;
FreeQ[{a,b,c,d},x] && FractionQ[n] && n<-1 && NonzeroQ[a*d+b*c]
```

$$\int \frac{(a + b\,x)^n\,(c + d\,x)^p}{x}\,dx$$

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{(a+b\,x)^n}{x} = b\,(a + b\,x)^{n-1} + \frac{a\,(a+b\,x)^{n-1}}{x}$
- **Rule: If** $n,\ p \in \mathbb{F}\ \land\ n > 0\ \land\ n - p \in \mathbb{Z}$, **then**

$$\int \frac{(a + b\,x)^n\,(c + d\,x)^p}{x}\,dx\ \longrightarrow\ b\int (a + b\,x)^{n-1}\,(c + d\,x)^p\,dx + a\int \frac{(a + b\,x)^{n-1}\,(c + d\,x)^p}{x}\,dx$$

- **Program code:**

```
Int[(a_+b_.*x_)^n_*(c_+d_.*x_)^p_/x_,x_Symbol] :=
  Dist[b,Int[(a+b*x)^(n-1)*(c+d*x)^p,x]] +
  Dist[a,Int[(a+b*x)^(n-1)*(c+d*x)^p/x,x]] /;
FreeQ[{a,b,c,d},x] && FractionQ[{n,p}] && n>0 && IntegerQ[n-p]
```

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{(a+b\,x)^n}{x} = \frac{(a+b\,x)^{n+1}}{a\,x} - \frac{b\,(a+b\,x)^n}{a}$
- **Rule: If** $n,\ p \in \mathbb{F}\ \land\ n < -1\ \land\ n - p \in \mathbb{Z}$, **then**

$$\int \frac{(a + b\,x)^n\,(c + d\,x)^p}{x}\,dx\ \longrightarrow\ \frac{1}{a}\int \frac{(a + b\,x)^{n+1}\,(c + d\,x)^p}{x}\,dx - \frac{b}{a}\int (a + b\,x)^n\,(c + d\,x)^p\,dx$$

- **Program code:**

```
Int[(a_+b_.*x_)^n_*(c_+d_.*x_)^p_/x_,x_Symbol] :=
  Dist[1/a,Int[(a+b*x)^(n+1)*(c+d*x)^p/x,x]] -
  Dist[b/a,Int[(a+b*x)^n*(c+d*x)^p,x]] /;
FreeQ[{a,b,c,d},x] && FractionQ[{n,p}] && n<-1 && IntegerQ[n-p]
```

$$\int x^m (a+b x)^n (c+d x)^n \, dx$$

- **Reference: G&R 2.174.2**

- **Rule: If $n \in \mathbb{F} \land m+2n+1=0 \land m>1 \land a d+b c=0$, then**

$$\int x^m (a+b x)^n (c+d x)^n \, dx \longrightarrow \frac{x^{m-1} (a+b x)^{n+1} (c+d x)^{n+1}}{2 b d (n+1)}+\frac{1}{b d} \int x^{m-2} (a+b x)^{n+1} (c+d x)^{n+1} \, dx$$

- **Program code:**

```
Int[x_^m_*(a_.+b_.*x_)^n_*(c_.+d_.*x_)^n_,x_Symbol] :=
  x^(m-1)*(a+b*x)^(n+1)*(c+d*x)^(n+1)/(2*b*d*(n+1)) +
  Dist[1/(b*d),Int[x^(m-2)*(a+b*x)^(n+1)*(c+d*x)^(n+1),x]] /;
FreeQ[{a,b,c,d},x] && FractionQ[n] && RationalQ[m] && m+2*n+1==0 && m>1 && ZeroQ[a*d+b*c]
```

- **Rule: If $n \in \mathbb{F} \land m+2n+1=0 \land m>1 \land a d+b c\neq 0$, then**

$$\int x^m (a+b x)^n (c+d x)^n \, dx \longrightarrow \frac{x^{m-1} (a+b x)^{n+1} (c+d x)^{n+1}}{2 b d (n+1)}-\frac{a d+b c}{2 b d} \int x^{m-1} (a+b x)^n (c+d x)^n \, dx+\frac{1}{b d} \int x^{m-2} (a+b x)^{n+1} (c+d x)^{n+1} \, dx$$

- **Program code:**

```
Int[x_^m_*(a_.+b_.*x_)^n_*(c_.+d_.*x_)^n_,x_Symbol] :=
  x^(m-1)*(a+b*x)^(n+1)*(c+d*x)^(n+1)/(2*b*d*(n+1)) -
  Dist[(a*d+b*c)/(2*b*d),Int[x^(m-1)*(a+b*x)^n*(c+d*x)^n,x]] +
  Dist[1/(b*d),Int[x^(m-2)*(a+b*x)^(n+1)*(c+d*x)^(n+1),x]] /;
FreeQ[{a,b,c,d},x] && FractionQ[n] && RationalQ[m] && m+2*n+1==0 && m>1 && NonzeroQ[a*d+b*c]
```

- **Reference: G&R 2.174.1, CRC 119**

- **Rule: If $n+1\neq 0 \land a d+b c=0$, then**

$$\int x (a+b x)^n (c+d x)^n \, dx \longrightarrow \frac{(a+b x)^{n+1} (c+d x)^{n+1}}{2 b d (n+1)}$$

- **Program code:**

```
Int[x_*(a_.+b_.*x_)^n_*(c_.+d_.*x_)^n_,x_Symbol] :=
  (a+b*x)^(n+1)*(c+d*x)^(n+1)/(2*b*d*(n+1)) /;
FreeQ[{a,b,c,d,n},x] && NonzeroQ[n+1] && ZeroQ[a*d+b*c]
```

- **Rule: If** $n \in \mathbb{F}$, **then**

$$\int x\,(a+b\,x)^n\,(c+d\,x)^n\,dx \longrightarrow \frac{(a+b\,x)^{n+1}\,(c+d\,x)^{n+1}}{2\,b\,d\,(n+1)} - \frac{a\,d+b\,c}{2\,b\,d}\int (a+b\,x)^n\,(c+d\,x)^n\,dx$$

- **Program code:**

```
Int[x_*(a_.+b_.*x_)^n_*(c_.+d_.*x_)^n_,x_Symbol] :=
  (a+b*x)^(n+1)*(c+d*x)^(n+1)/(2*b*d*(n+1)) -
  Dist[(a*d+b*c)/(2*b*d),Int[(a+b*x)^n*(c+d*x)^n,x]] /;
FreeQ[{a,b,c,d},x] && FractionQ[n]
```

- **Rule: If** $n \in \mathbb{F} \land m+2\,n+1 \neq 0 \land m>1 \land (m+n=0 \lor a\,d+b\,c=0)$, **then**

$$\int x^m\,(a+b\,x)^n\,(c+d\,x)^n\,dx \longrightarrow \frac{x^{m-1}\,(a+b\,x)^{n+1}\,(c+d\,x)^{n+1}}{b\,d\,(m+2\,n+1)} - \frac{a\,c\,(m-1)}{b\,d\,(m+2\,n+1)}\int x^{m-2}\,(a+b\,x)^n\,(c+d\,x)^n\,dx$$

- **Program code:**

```
Int[x_^m_*(a_.+b_.*x_)^n_*(c_.+d_.*x_)^n_,x_Symbol] :=
  x^(m-1)*(a+b*x)^(n+1)*(c+d*x)^(n+1)/(b*d*(m+2*n+1)) -
  Dist[a*c*(m-1)/(b*d*(m+2*n+1)),Int[x^(m-2)*(a+b*x)^n*(c+d*x)^n,x]] /;
FreeQ[{a,b,c,d},x] && FractionQ[n] && RationalQ[m] && NonzeroQ[m+2*n+1] && m>1 &&
(ZeroQ[m+n] || ZeroQ[a*d+b*c])
```

- **Rule: If** $n \in \mathbb{F} \land m+2\,n+1 \neq 0 \land m>1 \land m+n \neq 0 \land a\,d+b\,c \neq 0$, **then**

$$\int x^m\,(a+b\,x)^n\,(c+d\,x)^n\,dx \longrightarrow \frac{x^{m-1}\,(a+b\,x)^{n+1}\,(c+d\,x)^{n+1}}{b\,d\,(m+2\,n+1)} - \frac{(m+n)\,(a\,d+b\,c)}{b\,d\,(m+2\,n+1)}\int x^{m-1}\,(a+b\,x)^n\,(c+d\,x)^n\,dx - \frac{a\,c\,(m-1)}{b\,d\,(m+2\,n+1)}\int x^{m-2}\,(a+b\,x)^n\,(c+d\,x)^n\,dx$$

- **Program code:**

```
Int[x_^m_*(a_.+b_.*x_)^n_*(c_.+d_.*x_)^n_,x_Symbol] :=
  x^(m-1)*(a+b*x)^(n+1)*(c+d*x)^(n+1)/(b*d*(m+2*n+1)) -
  Dist[(m+n)*(a*d+b*c)/(b*d*(m+2*n+1)),Int[x^(m-1)*(a+b*x)^n*(c+d*x)^n,x]] -
  Dist[a*c*(m-1)/(b*d*(m+2*n+1)),Int[x^(m-2)*(a+b*x)^n*(c+d*x)^n,x]] /;
FreeQ[{a,b,c,d},x] && FractionQ[n] && RationalQ[m] && NonzeroQ[m+2*n+1] && m>1 &&
NonzeroQ[m+n] && NonzeroQ[a*d+b*c]
```

- **Reference: G&R 2.176, CRC 123**
- **Rule: If** $m+1\neq 0\land m+2n+3=0\land a d+b c=0$**, then**

$$\int x^m (a+b x)^n (c+d x)^n\,dx\longrightarrow\frac{x^{m+1}(a+b x)^{n+1}(c+d x)^{n+1}}{a c(m+1)}$$

- **Program code:**

```
Int[x_^m_*(a_+b_.*x_)^n_*(c_+d_.*x_)^n_,x_Symbol] :=
  x^(m+1)*(a+b*x)^(n+1)*(c+d*x)^(n+1)/(a*c*(m+1)) /;
FreeQ[{a,b,c,d,n},x] && NonzeroQ[m+1] && ZeroQ[m+2*n+3] && ZeroQ[a*d+b*c]
```

- **Rule: If** $n\in\mathbb{F}\land m<-1\land m+2n+3=0$**, then**

$$\int x^m (a+b x)^n (c+d x)^n\,dx\longrightarrow$$
$$\frac{x^{m+1}(a+b x)^{n+1}(c+d x)^{n+1}}{a c(m+1)}-\frac{(m+n+2)(a d+b c)}{a c(m+1)}\int x^{m+1}(a+b x)^n (c+d x)^n\,dx$$

- **Program code:**

```
Int[x_^m_*(a_+b_.*x_)^n_*(c_+d_.*x_)^n_,x_Symbol] :=
  x^(m+1)*(a+b*x)^(n+1)*(c+d*x)^(n+1)/(a*c*(m+1)) -
  Dist[(m+n+2)*(a*d+b*c)/(a*c*(m+1)),Int[x^(m+1)*(a+b*x)^n*(c+d*x)^n,x]] /;
FreeQ[{a,b,c,d},x] && FractionQ[n] && RationalQ[m] && m<-1 && ZeroQ[m+2*n+3]
```

- **Rule: If** $n\in\mathbb{F}\land m<-1\land(m+n+2=0\lor a d+b c=0)$**, then**

$$\int x^m (a+b x)^n (c+d x)^n\,dx\longrightarrow\frac{x^{m+1}(a+b x)^{n+1}(c+d x)^{n+1}}{a c(m+1)}-\frac{b d(m+2n+3)}{a c(m+1)}\int x^{m+2}(a+b x)^n (c+d x)^n\,dx$$

- **Program code:**

```
Int[x_^m_*(a_+b_.*x_)^n_*(c_+d_.*x_)^n_,x_Symbol] :=
  x^(m+1)*(a+b*x)^(n+1)*(c+d*x)^(n+1)/(a*c*(m+1)) -
  Dist[b*d*(m+2*n+3)/(a*c*(m+1)),Int[x^(m+2)*(a+b*x)^n*(c+d*x)^n,x]] /;
FreeQ[{a,b,c,d},x] && FractionQ[n] && RationalQ[m] && m<-1 && (ZeroQ[m+n+2] || ZeroQ[a*d+b*c])
```

- **Rule: If** $n \in \mathbb{F} \land m < -1 \land m + n + 2 \neq 0 \land m + 2n + 3 \neq 0 \land a d + b c \neq 0$**, then**

$$\int x^m (a + b x)^n (c + d x)^n \, dx \longrightarrow \frac{x^{m+1} (a + b x)^{n+1} (c + d x)^{n+1}}{a c (m + 1)} - \frac{(m + n + 2) (a d + b c)}{a c (m + 1)} \int x^{m+1} (a + b x)^n (c + d x)^n \, dx - \frac{b d (m + 2 n + 3)}{a c (m + 1)} \int x^{m+2} (a + b x)^n (c + d x)^n \, dx$$

- **Program code:**

```
Int[x_^m_*(a_+b_.*x_)^n_*(c_+d_.*x_)^n_,x_Symbol] :=
  x^(m+1)*(a+b*x)^(n+1)*(c+d*x)^(n+1)/(a*c*(m+1)) -
  Dist[(m+n+2)*(a*d+b*c)/(a*c*(m+1)),Int[x^(m+1)*(a+b*x)^n*(c+d*x)^n,x]] -
  Dist[b*d*(m+2*n+3)/(a*c*(m+1)),Int[x^(m+2)*(a+b*x)^n*(c+d*x)^n,x]] /;
FreeQ[{a,b,c,d},x] && FractionQ[n] && RationalQ[m] && m<-1 && NonzeroQ[m+n+2] &&
NonzeroQ[m+2*n+3] && NonzeroQ[a*d+b*c]
```

$$\int x^m (a+b x)^n (c+d x)^p \, dx$$

- **Rule: If $m \in \mathbb{Z} \land m > 0$, then**

$$\int \frac{x^m \sqrt{a+b x}}{\sqrt{c+d x}} \, dx \longrightarrow \frac{x^m \sqrt{a+b x} \sqrt{c+d x}}{d (m+1)} - \frac{a c m}{d (m+1)} \int \frac{x^{m-1}}{\sqrt{a+b x} \sqrt{c+d x}} \, dx + \frac{a d - b c (2 m+1)}{2 d (m+1)} \int \frac{x^m}{\sqrt{a+b x} \sqrt{c+d x}} \, dx$$

- **Program code:**

```
Int[x_^m_.*Sqrt[a_+b_.*x_]/Sqrt[c_.+d_.*x_],x_Symbol] :=
  x^m*Sqrt[a+b*x]*Sqrt[c+d*x]/(d*(m+1)) -
  Dist[a*c*m/(d*(m+1)),Int[x^(m-1)/(Sqrt[a+b*x]*Sqrt[c+d*x]),x]] +
  Dist[(a*d-b*c*(2*m+1))/(2*d*(m+1)),Int[x^m/(Sqrt[a+b*x]*Sqrt[c+d*x]),x]] /;
FreeQ[{a,b,c,d},x] && IntegerQ[m] && m>0
```

- **Rule:**

$$\int \frac{\sqrt{a+b x}}{x^2 \sqrt{c+d x}} \, dx \longrightarrow -\frac{\sqrt{a+b x} \sqrt{c+d x}}{c x} + \frac{b c - a d}{2 c} \int \frac{1}{x \sqrt{a+b x} \sqrt{c+d x}} \, dx$$

- **Program code:**

```
Int[Sqrt[a_+b_.*x_]/(x_^2*Sqrt[c_.+d_.*x_]),x_Symbol] :=
  -Sqrt[a+b*x]*Sqrt[c+d*x]/(c*x) +
  Dist[(b*c-a*d)/(2*c),Int[1/(x*Sqrt[a+b*x]*Sqrt[c+d*x]),x]] /;
FreeQ[{a,b,c,d},x]
```

- **Rule: If** $m \in \mathbb{Z} \wedge m < -2$, **then**

$$\int \frac{x^m \sqrt{a+b x}}{\sqrt{c+d x}} \, dx \longrightarrow \frac{x^{m+1} \sqrt{a+b x} \sqrt{c+d x}}{c (m+1)} - \frac{b c+a d (2 m+3)}{2 c (m+1)} \int \frac{x^{m+1}}{\sqrt{a+b x} \sqrt{c+d x}} \, dx - \frac{b d (m+2)}{c (m+1)} \int \frac{x^{m+2}}{\sqrt{a+b x} \sqrt{c+d x}} \, dx$$

- **Program code:**

```
Int[x_^m_.*Sqrt[a_+b_.*x_]/Sqrt[c_.+d_.*x_],x_Symbol] :=
  x^(m+1)*Sqrt[a+b*x]*Sqrt[c+d*x]/(c*(m+1)) -
  Dist[(b*c+a*d*(2*m+3))/(2*c*(m+1)),Int[x^(m+1)/(Sqrt[a+b*x]*Sqrt[c+d*x]),x]] -
  Dist[b*d*(m+2)/(c*(m+1)),Int[x^(m+2)/(Sqrt[a+b*x]*Sqrt[c+d*x]),x]] /;
FreeQ[{a,b,c,d},x] && IntegerQ[m] && m<-2
```

- **Derivation: Algebraic expansion**
- **Basis:** $x^m (a+b x)^n = \frac{x^{m-1} (a+b x)^{n+1}}{b} - \frac{a x^{m-1} (a+b x)^n}{b}$
- **Basis: If** $m \geq 0$ **is an integer, then** $x^m = \sum_{k=0}^{m} \frac{(-a)^{m-k} \text{Binomial}[m,m-k]}{b^m} (a+b x)^k$
- **Rule: If** $m, p-n \in \mathbb{Z} \bigwedge m > 0 \bigwedge p-n < 0 \bigwedge \left(m > 3 \bigvee n \neq -\frac{1}{2}\right)$, **then**

$$\int x^m (a+b x)^n (c+d x)^p \, dx \longrightarrow \sum_{k=0}^{m} \frac{(-a)^{m-k} \text{Binomial}[m, m-k]}{b^m} \int (a+b x)^{n+k} (c+d x)^p \, dx$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*x_)^n_*(c_.+d_.*x_)^p_.,x_Symbol] :=
  Sum[Dist[(-a)^(m-k)/b^m*Binomial[m,m-k],Int[(a+b*x)^(n+k)*(c+d*x)^p,x]],{k,0,m}] /;
FreeQ[{a,b,c,d,n,p},x] && IntegersQ[m,p-n] && m>0 && Not[IntegerQ[n]] && p-n<0 &&
(m>3 || n=!=-1/2)
```

- **Derivation: Algebraic expansion**

- **Basis:** $x^m (a+b x)^n = \frac{x^{m-1} (a+b x)^{n+1}}{b} - \frac{a x^{m-1} (a+b x)^n}{b}$

- **Basis: If** m **and** $p-n$ **are integers and** $0 < p-n \leq m$**, then**

$$x^m (a+b x)^n = \sum_{k=0}^{p-n-1} \frac{(-a)^{m-k} \text{Binomial}[m, m-k]}{b^m} (a+b x)^{n+k} + \sum_{k=0}^{m-p+n} \frac{\left(-\frac{a}{b}\right)^{m-k} \text{Binomial}[m-k-1, p-n-1]}{(-a)^{p-n}} x^k (a+b x)^p$$

- **Rule: If** $m,\ p-n \in \mathbb{Z} \land 0 < p-n \leq m$**, then**

$$\int x^m (a+b x)^n (c+d x)^p \, dx \longrightarrow \sum_{k=0}^{p-n-1} \frac{(-a)^{m-k} \text{Binomial}[m, m-k]}{b^m} \int (a+b x)^{n+k} (c+d x)^p \, dx +$$
$$\sum_{k=0}^{m-p+n} \frac{\left(-\frac{a}{b}\right)^{m-k} \text{Binomial}[m-k-1, p-n-1]}{(-a)^{p-n}} \int x^k (a+b x)^p (c+d x)^p \, dx$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*x_)^n_*(c_.+d_.*x_)^p_,x_Symbol] :=
  Sum[Dist[(-a)^(m-k)/b^m*Binomial[m,m-k],Int[(a+b*x)^(n+k)*(c+d*x)^p,x]],{k,0,p-n-1}] +
  Sum[Dist[(-a/b)^(m-k)/(-a)^(p-n)*Binomial[m-k-1,p-n-1],Int[x^k*(a+b*x)^p*(c+d*x)^p,x]],{k,0,m-p+n}
FreeQ[{a,b,c,d,n,p},x] && IntegersQ[m,p-n] && 0<p-n<=m && Not[IntegerQ[n]]
```

- **Derivation: Algebraic expansion**

- **Basis:** $x^m (a+b x)^n = \frac{x^m (a+b x)^{n+1}}{a} - \frac{b x^{m+1} (a+b x)^n}{a}$

- **Basis: If** m **and** $p-n$ **are integers,** $m < 0$ **and** $p-n > 0$**, then**

$$x^m (a+b x)^n = \sum_{k=0}^{p-n-1} \frac{a^{m-k} \text{Binomial}[k-m-1, -m-1]}{(-b)^m} (a+b x)^{n+k} + \sum_{k=0}^{-m-1} \frac{\left(-\frac{b}{a}\right)^k \text{Binomial}[p-n+k-1, p-n-1]}{a^{p-n}} x^{m+k} (a+b x)^p$$

- **Rule: If** $m,\ p-n \in \mathbb{Z} \land m < 0 \land p-n > 0$**, then**

$$\int x^m (a+b x)^n (c+d x)^p \, dx \longrightarrow \sum_{k=0}^{p-n-1} \frac{a^{m-k} \text{Binomial}[k-m-1, -m-1]}{(-b)^m} \int (a+b x)^{n+k} (c+d x)^p \, dx +$$
$$\sum_{k=0}^{-m-1} \frac{\left(-\frac{b}{a}\right)^k \text{Binomial}[p-n+k-1, p-n-1]}{a^{p-n}} \int x^{m+k} (a+b x)^p (c+d x)^p \, dx$$

- **Program code:**

```
Int[x_^m_.*(a_+b_.*x_)^n_*(c_.+d_.*x_)^p_,x_Symbol] :=
  Sum[Dist[a^(m-k)/(-b)^m*Binomial[k-m-1,-m-1],Int[(a+b*x)^(n+k)*(c+d*x)^p,x]],{k,0,p-n-1}] +
  Sum[Dist[(-b/a)^k/a^(p-n)*Binomial[p-n+k-1,p-n-1],Int[x^(m+k)*(a+b*x)^p*(c+d*x)^p,x]],{k,0,-m-1}]
FreeQ[{a,b,c,d,n,p},x] && IntegersQ[m,p-n] && m<0 && p-n>0 && Not[IntegerQ[n]]
```