$$\int \text{ArcCsch}[a+b\,x]^n \, dx$$

- **Reference: CRC 594', A&S 4.6.46'**
- **Derivation: Integration by parts**
- **Rule:**

$$\int \text{ArcCsch}[a+b\,x]\, dx \longrightarrow \frac{(a+b\,x)\,\text{ArcCsch}[a+b\,x]}{b} + \frac{1}{b}\,\text{ArcTanh}\left[\sqrt{1+\frac{1}{(a+b\,x)^2}}\,\right]$$

- **Program code:**

```
Int[ArcCsch[a_.+b_.*x_],x_Symbol] :=
  (a+b*x)*ArcCsch[a+b*x]/b + ArcTanh[Sqrt[1+1/(a+b*x)^2]]/b /;
FreeQ[{a,b},x]
```

$$\int x^m \,\text{ArcCsch}[a+b\,x]\,dx$$

- **Derivation: Integration by substitution**
- **Rule: If** $m \in \mathbb{Z} \wedge m > 0$**, then**

$$\int x^m \,\text{ArcCsch}[a+b\,x]\,dx \longrightarrow \frac{1}{b}\,\text{Subst}\left[\int \left(-\frac{a}{b}+\frac{x}{b}\right)^m \text{ArcCsch}[x]\,dx,\ x,\ a+b\,x\right]$$

- **Program code:**

```
Int[x_^m_.*ArcCsch[a_+b_.*x_],x_Symbol] :=
  Dist[1/b,Subst[Int[(-a/b+x/b)^m*ArcCsch[x],x],x,a+b*x]] /;
FreeQ[{a,b},x] && IntegerQ[m] && m>0
```

- **Reference: CRC 596, A&S 4.6.56**
- **Derivation: Integration by parts**
- **Rule: If** $m+1 \neq 0$**, then**

$$\int x^m \,\text{ArcCsch}[a+b\,x]\,dx \longrightarrow \frac{x^{m+1}\,\text{ArcCsch}[a+b\,x]}{m+1}+\frac{b}{m+1}\int \frac{x^{m+1}}{(a+b\,x)^2\sqrt{1+\frac{1}{(a+b\,x)^2}}}\,dx$$

- **Program code:**

```
Int[x_^m_.*ArcCsch[a_.+b_.*x_],x_Symbol] :=
  x^(m+1)*ArcCsch[a+b*x]/(m+1) +
  Dist[b/(m+1),Int[x^(m+1)/((a+b*x)^2*Sqrt[1+1/(a+b*x)^2]),x]] /;
FreeQ[{a,b,m},x] && NonzeroQ[m+1]
```

```
(* Int[ArcCsch[a_.*x_^n_.]/x_,x_Symbol] :=
(* Int[ArcCsch[1/a*x^(-n)]/x,x] /; *)
  -ArcCsch[a*x^n]^2/(2*n) -
  ArcCsch[a*x^n]*Log[1-E^(-2*ArcCsch[a*x^n])]/n +
  PolyLog[2,E^(-2*ArcCsch[a*x^n])]/(2*n) /;
(* -ArcCsch[a*x^n]^2/(2*n) -
  ArcCsch[a*x^n]*Log[1-1/(1/(a*x^n)+Sqrt[1+1/(a^2*x^(2*n))])^2]/n +
  PolyLog[2,1/(1/(a*x^n)+Sqrt[1+1/(a^2*x^(2*n))])^2]/(2*n) /; *)
FreeQ[{a,n},x] *)
```

$$\int u\,\text{ArcCsch}\left[\frac{c}{a+b\,x^n}\right]^m dx$$

- **Derivation: Algebraic simplification**
- **Basis:** $\text{ArcCsch}[z] = \text{ArcSinh}\left[\frac{1}{z}\right]$
- **Rule:**

$$\int u\,\text{ArcCsch}\left[\frac{c}{a+b\,x^n}\right]^m dx \longrightarrow \int u\,\text{ArcSinh}\left[\frac{a}{c}+\frac{b\,x^n}{c}\right]^m dx$$

- **Program code:**

```
Int[u_.*ArcCsch[c_./(a_.+b_.*x_^n_.)]^m_.,x_Symbol] :=
  Int[u*ArcSinh[a/c+b*x^n/c]^m,x] /;
FreeQ[{a,b,c,n,m},x]
```

$$\int \text{ArcCsch}[u]\, dx$$

- **Derivation: Integration by parts**

- **Rule: If u is free of inverse functions, then**

$$\int \text{ArcCsch}[u]\, dx \longrightarrow x\, \text{ArcCsch}[u] + \int \frac{x\, \partial_x u}{u^2 \sqrt{1 + \frac{1}{u^2}}}\, dx$$

- **Program code:**

```
Int[ArcCsch[u_],x_Symbol] :=
  x*ArcCsch[u] +
  Int[Regularize[x*D[u,x]/(u^2*Sqrt[1+1/u^2]),x],x] /;
InverseFunctionFreeQ[u,x] && Not[FunctionOfExponentialOfLinear[u,x]]
```

$\int x^m e^{n\,\text{ArcCsch}[u]}\,dx$

- **Derivation: Algebraic simplification**

- **Basis:** $e^{n\,\text{ArcCsch}[z]} = \left(\frac{1}{z} + \sqrt{1 + \frac{1}{z^2}}\right)^n$

- **Rule: If** $n \in \mathbb{Z} \land u$ **is a polynomial in** x**, then**

$$\int e^{n\,\text{ArcCsch}[u]}\,dx \longrightarrow \int \left(\frac{1}{u} + \sqrt{1 + \frac{1}{u^2}}\right)^n dx$$

- **Program code:**

```
Int[E^(n_.*ArcCsch[u_]), x_Symbol] :=
  Int[(1/u+Sqrt[1+1/u^2])^n,x] /;
IntegerQ[n] && PolynomialQ[u,x]
```

- **Derivation: Algebraic simplification**

- **Basis:** $e^{n\,\text{ArcCsch}[z]} = \left(\frac{1}{z} + \sqrt{1 + \frac{1}{z^2}}\right)^n$

- **Rule: If** $n \in \mathbb{Z} \land u$ **is a polynomial in** x**, then**

$$\int x^m e^{n\,\text{ArcCsch}[u]}\,dx \longrightarrow \int x^m \left(\frac{1}{u} + \sqrt{1 + \frac{1}{u^2}}\right)^n dx$$

- **Program code:**

```
Int[x_^m_.*E^(n_.*ArcCsch[u_]), x_Symbol] :=
  Int[x^m*(1/u+Sqrt[1+1/u^2])^n,x] /;
RationalQ[m] && IntegerQ[n] && PolynomialQ[u,x]
```