$$\int \mathrm{ArcSech}[a+b\,x]^n\,dx$$

- **Reference: CRC 591', A&S 4.6.47'**
- **Derivation: Integration by parts**
- **Rule:**

$$\int \mathrm{ArcSech}[a+b\,x]\,dx \longrightarrow \frac{(a+b\,x)\,\mathrm{ArcSech}[a+b\,x]}{b}-\frac{2}{b}\,\mathrm{ArcTan}\left[\sqrt{\frac{1-a-b\,x}{1+a+b\,x}}\right]$$

- **Program code:**

```
Int[ArcSech[a_.+b_.*x_],x_Symbol] :=
  (a+b*x)*ArcSech[a+b*x]/b - 2*ArcTan[Sqrt[(1-a-b*x)/(1+a+b*x)]]/b /;
FreeQ[{a,b},x]
```

# $\int x^m \operatorname{ArcSech}[a+b\,x]\,dx$

- **Derivation: Integration by substitution**
- **Rule: If** $m \in \mathbb{Z} \land m > 0$, **then**

$$\int x^m \operatorname{ArcSech}[a+b\,x]\,dx \longrightarrow \frac{1}{b}\,\text{Subst}\left[\int\left(-\frac{a}{b}+\frac{x}{b}\right)^m \operatorname{ArcSech}[x]\,dx,\ x,\ a+b\,x\right]$$

- **Program code:**

```
Int[x_^m_.*ArcSech[a_+b_.*x_],x_Symbol] :=
  Dist[1/b,Subst[Int[(-a/b+x/b)^m*ArcSech[x],x],x,a+b*x]] /;
FreeQ[{a,b},x] && IntegerQ[m] && m>0
```

- **Reference: CRC 593', A&S 4.6.58'**
- **Derivation: Integration by parts**
- **Basis:** $\partial_x \operatorname{ArcSech}[x] = -\frac{\sqrt{\frac{1}{1+x}}\,\sqrt{1+x}}{x\,\sqrt{1+x}\,\sqrt{1-x}}$
- **Basis:** $\partial_x\left(\sqrt{\frac{1}{1+a+b\,x^n}}\,\sqrt{1+a+b\,x^n}\right) = 0$
- **Rule: If** $m+1 \neq 0$, **then**

$$\int x^m \operatorname{ArcSech}[a\,x]\,dx \longrightarrow \frac{x^{m+1}\operatorname{ArcSech}[a\,x]}{m+1}+\frac{1}{m+1}\int\frac{x^m}{\sqrt{\frac{1-a\,x}{1+a\,x}}\,(1+a\,x)}\,dx$$

- **Program code:**

```
Int[x_^m_.*ArcSech[a_.*x_],x_Symbol] :=
  x^(m+1)*ArcSech[a*x]/(m+1) +
  Dist[1/(m+1),Int[x^m/(Sqrt[(1-a*x)/(1+a*x)]*(1+a*x)),x]] /;
FreeQ[{a,m},x] && NonzeroQ[m+1]
```

```
(* Int[ArcSech[a_.*x_^n_.]/x_,x_Symbol] :=
(*   Int[ArcCosh[1/a*x^(-n)]/x,x] /; *)
  -ArcSech[a*x^n]^2/(2*n) -
  ArcSech[a*x^n]*Log[1+E^(-2*ArcSech[a*x^n])]/n +
  PolyLog[2,-E^(-2*ArcSech[a*x^n])]/(2*n) /;
(* -ArcSech[a*x^n]^2/(2*n) -
  ArcSech[a*x^n]*Log[1+1/(1/(a*x^n)+Sqrt[-1+1/(a*x^n)]*Sqrt[1+1/(a*x^n)])^2]/n +
  PolyLog[2,-1/(1/(a*x^n)+Sqrt[-1+1/(a*x^n)]*Sqrt[1+1/(a*x^n)])^2]/(2*n) /; *)
FreeQ[{a,n},x] *)
```

$$\int u \operatorname{ArcSech}\left[\frac{c}{a+b x^n}\right]^m dx$$

- **Derivation: Algebraic simplification**
- **Basis:** $\operatorname{ArcSech}[z] = \operatorname{ArcCosh}\left[\frac{1}{z}\right]$
- **Rule:**

$$\int u \operatorname{ArcSech}\left[\frac{c}{a+b x^n}\right]^m dx \longrightarrow \int u \operatorname{ArcCosh}\left[\frac{a}{c}+\frac{b x^n}{c}\right]^m dx$$

- **Program code:**

```
Int[u_.*ArcSech[c_./(a_.+b_.*x_^n_.)]^m_.,x_Symbol] :=
  Int[u*ArcCosh[a/c+b*x^n/c]^m,x] /;
FreeQ[{a,b,c,n,m},x]
```

$$\int \text{ArcSech}[u]\, dx$$

- **Derivation: Integration by parts**

- **Rule: If u is free of inverse functions, then**

$$\int \text{ArcSech}[u]\, dx \longrightarrow x\,\text{ArcSech}[u] + \int \frac{x\,\partial_x u}{u^2 \sqrt{-1+\frac{1}{u}}\,\sqrt{1+\frac{1}{u}}}\, dx$$

- **Program code:**

```
(* Int[ArcSech[u_],x_Symbol] :=
  x*ArcSech[u] +
  Int[Regularize[x*D[u,x]/(u^2*Sqrt[-1+1/u]*Sqrt[1+1/u]),x],x] /;
InverseFunctionFreeQ[u,x] && Not[FunctionOfExponentialOfLinear[u,x]] *)
```

# $\int x^m e^{n \text{ArcSech}[u]} \, dx$

- **Derivation: Algebraic simplification**
- **Basis:** $e^{n \text{ArcSech}[z]} = \left(\sqrt{-1+\frac{1}{z}}\,\sqrt{1+\frac{1}{z}}+\frac{1}{z}\right)^n$
- **Basis: If** $n \in \mathbb{Z}$, $e^{n \text{ArcSech}[z]} = \left(\frac{1}{z}+\sqrt{\frac{1-z}{1+z}}+\frac{\sqrt{\frac{1-z}{1+z}}}{z}\right)^n$
- **Basis: If** $n \in \mathbb{Z}$, $e^{n \text{ArcSech}[z]} = \left(\frac{1+\frac{\sqrt{1-z}}{\sqrt{\frac{1}{1+z}}}}{z}\right)^n$
- **Rule: If** $n \in \mathbb{Z} \wedge u$ **is a polynomial in** $x$**, then**

$$\int e^{n \text{ArcSech}[u]} \, dx \longrightarrow \int \left(\frac{1}{u}+\sqrt{\frac{1-u}{1+u}}+\frac{\sqrt{\frac{1-u}{1+u}}}{u}\right)^n dx$$

- **Program code:**

```
Int[E^(n_.*ArcSech[u_]), x_Symbol] :=
  Int[(1/u + Sqrt[(1-u)/(1+u)] + Sqrt[(1-u)/(1+u)]/u)^n,x] /;
IntegerQ[n] && PolynomialQ[u,x]
```

- **Derivation: Algebraic simplification**
- **Basis:** $e^{n \text{ArcSech}[z]} = \left(\sqrt{-1+\frac{1}{z}}\,\sqrt{1+\frac{1}{z}}+\frac{1}{z}\right)^n$
- **Rule: If** $n \in \mathbb{Z} \wedge u$ **is a polynomial in** $x$**, then**

$$\int x^m e^{n \text{ArcSech}[u]} \, dx \longrightarrow \int x^m \left(\frac{1}{u}+\sqrt{\frac{1-u}{1+u}}+\frac{\sqrt{\frac{1-u}{1+u}}}{u}\right)^n dx$$

- **Program code:**

```
Int[x_^m_.*E^(n_.*ArcSech[u_]), x_Symbol] :=
  Int[x^m*(1/u + Sqrt[(1-u)/(1+u)] + Sqrt[(1-u)/(1+u)]/u)^n,x] /;
RationalQ[m] && IntegerQ[n] && PolynomialQ[u,x]
```