$\int \text{ExpIntegralE}[n, a+b\,x]\,dx$

- **Basis:** $\frac{\partial E_n(z)}{\partial z} = -E_{n-1}(z)$
- **Rule:**

$$\int \text{ExpIntegralE}[n, a+b\,x]\,dx \rightarrow -\frac{\text{ExpIntegralE}[n+1, a+b\,x]}{b}$$

- **Program code:**

```
Int[ExpIntegralE[n_,a_.+b_.*x_],x_Symbol] :=
  -ExpIntegralE[n+1,a+b*x]/b /;
FreeQ[{a,b,n},x]
```

$$\int x^m \text{ExpIntegralE}[n, a+b x] \, dx$$

- **Derivation: Integration by parts**
- **Rule: If $n \in \mathbb{Z} \wedge n > 0$, then**

$$\int x^m \text{ExpIntegralE}[n, a+b x] \, dx \longrightarrow \frac{x^{m+1} \text{ExpIntegralE}[n, a+b x]}{m+1} + \frac{b}{m+1} \int x^{m+1} \text{ExpIntegralE}[n-1, a+b x] \, dx$$

- **Program code:**

```
Int[x_^m_.*ExpIntegralE[n_,a_.+b_.*x_],x_Symbol] :=
  x^(m+1)*ExpIntegralE[n,a+b*x]/(m+1) +
  Dist[b/(m+1),Int[x^(m+1)*ExpIntegralE[n-1,a+b*x],x]] /;
FreeQ[{a,b,m},x] && IntegerQ[n] && n>0
```

- **Derivation: Inverted integration by parts**
- **Rule: If $n \in \mathbb{Z} \wedge n > 0$, then**

$$\int x^m \text{ExpIntegralE}[n, a+b x] \, dx \longrightarrow -\frac{x^m \text{ExpIntegralE}[n+1, a+b x]}{b} + \frac{m}{b} \int x^{m-1} \text{ExpIntegralE}[n+1, a+b x] \, dx$$

- **Program code:**

```
Int[x_^m_.*ExpIntegralE[n_,a_.+b_.*x_],x_Symbol] :=
  -x^m*ExpIntegralE[n+1,a+b*x]/b +
  Dist[m/b,Int[x^(m-1)*ExpIntegralE[n+1,a+b*x],x]] /;
FreeQ[{a,b,m},x] && IntegerQ[n] && n<0
```

$$\int \text{ExpIntegralEi}[a + b\,x]^n \, dx$$

- **Derivation: Integration by parts**
- **Rule:**

$$\int \text{ExpIntegralEi}[a + b\,x] \, dx \longrightarrow \frac{(a + b\,x)\,\text{ExpIntegralEi}[a + b\,x]}{b} - \frac{e^{a+b\,x}}{b}$$

- **Program code:**

```
Int[ExpIntegralEi[a_.+b_.*x_],x_Symbol] :=
  (a+b*x)*ExpIntegralEi[a+b*x]/b - E^(a+b*x)/b /;
FreeQ[{a,b},x]
```

- **Derivation: Integration by parts**
- **Rule:**

$$\int \text{ExpIntegralEi}[a + b\,x]^2 \, dx \longrightarrow \frac{(a + b\,x)\,\text{ExpIntegralEi}[a + b\,x]^2}{b} - 2\int e^{a+b\,x}\,\text{ExpIntegralEi}[a + b\,x] \, dx$$

- **Program code:**

```
Int[ExpIntegralEi[a_.+b_.*x_]^2,x_Symbol] :=
  (a+b*x)*ExpIntegralEi[a+b*x]^2/b -
  Dist[2,Int[E^(a+b*x)*ExpIntegralEi[a+b*x],x]] /;
FreeQ[{a,b},x]
```

$$\int x^m \operatorname{ExpIntegralEi}[a+b\,x]^n\,dx$$

- **Derivation: Integration by parts**
- **Rule: If $m+1 \neq 0$, then**

$$\int x^m \operatorname{ExpIntegralEi}[a+b\,x]\,dx \longrightarrow \frac{x^{m+1} \operatorname{ExpIntegralEi}[a+b\,x]}{m+1} - \frac{b}{m+1}\int \frac{x^{m+1}\,e^{a+b\,x}}{a+b\,x}\,dx$$

- **Program code:**

```
Int[x_^m_.*ExpIntegralEi[a_.+b_.*x_],x_Symbol] :=
  x^(m+1)*ExpIntegralEi[a+b*x]/(m+1) -
  Dist[b/(m+1),Int[x^(m+1)*E^(a+b*x)/(a+b*x),x]] /;
FreeQ[{a,b,m},x] && NonzeroQ[m+1]
```

- **Derivation: Integration by parts**
- **Rule: If $m \in \mathbb{Z} \wedge m > 0$, then**

$$\int x^m \operatorname{ExpIntegralEi}[b\,x]^2\,dx \longrightarrow \frac{x^{m+1} \operatorname{ExpIntegralEi}[b\,x]^2}{m+1} - \frac{2}{m+1}\int x^m\,e^{b\,x} \operatorname{ExpIntegralEi}[b\,x]\,dx$$

- **Program code:**

```
Int[x_^m_.*ExpIntegralEi[b_.*x_]^2,x_Symbol] :=
  x^(m+1)*ExpIntegralEi[b*x]^2/(m+1) -
  Dist[2/(m+1),Int[x^m*E^(b*x)*ExpIntegralEi[b*x],x]] /;
FreeQ[b,x] && IntegerQ[m] && m>0
```

- **Derivation: Iterated integration by parts**
- **Rule: If $m \in \mathbb{Z} \wedge m > 0$, then**

$$\int x^m \operatorname{ExpIntegralEi}[a+b\,x]^2\,dx \longrightarrow \frac{x^{m+1} \operatorname{ExpIntegralEi}[a+b\,x]^2}{m+1} + \frac{a\,x^m \operatorname{ExpIntegralEi}[a+b\,x]^2}{b\,(m+1)} - \frac{2}{m+1}\int x^m\,e^{a+b\,x} \operatorname{ExpIntegralEi}[a+b\,x]\,dx - \frac{a\,m}{b\,(m+1)}\int x^{m-1} \operatorname{ExpIntegralEi}[a+b\,x]^2\,dx$$

- **Program code:**

```
Int[x_^m_.*ExpIntegralEi[a_+b_.*x_]^2,x_Symbol] :=
  x^(m+1)*ExpIntegralEi[a+b*x]^2/(m+1) +
  a*x^m*ExpIntegralEi[a+b*x]^2/(b*(m+1)) -
  Dist[2/(m+1),Int[x^m*E^(a+b*x)*ExpIntegralEi[a+b*x],x]] -
  Dist[a*m/(b*(m+1)),Int[x^(m-1)*ExpIntegralEi[a+b*x]^2,x]] /;
FreeQ[{a,b},x] && IntegerQ[m] && m>0
```

- **Derivation: Inverted integration by parts**

- **Rule: If $m \in \mathbb{Z} \land m < -2$, then**

$$\int x^m \operatorname{ExpIntegralEi}[a+b\,x]^2\,dx \longrightarrow \frac{b\,x^{m+2}\operatorname{ExpIntegralEi}[a+b\,x]^2}{a\,(m+1)}+\frac{x^{m+1}\operatorname{ExpIntegralEi}[a+b\,x]^2}{m+1}-\frac{2\,b}{a\,(m+1)}\int x^{m+1}\,e^{a+b\,x}\operatorname{ExpIntegralEi}[a+b\,x]\,dx-\frac{b\,(m+2)}{a\,(m+1)}\int x^{m+1}\operatorname{ExpIntegralEi}[a+b\,x]^2\,dx$$

- **Program code:**

```
(* Int[x_^m_.*ExpIntegralEi[a_+b_.*x_]^2,x_Symbol] :=
  b*x^(m+2)*ExpIntegralEi[a+b*x]^2/(a*(m+1)) +
  x^(m+1)*ExpIntegralEi[a+b*x]^2/(m+1) -
  Dist[2*b/(a*(m+1)),Int[x^(m+1)*E^(a+b*x)*ExpIntegralEi[a+b*x],x]] -
  Dist[b*(m+2)/(a*(m+1)),Int[x^(m+1)*ExpIntegralEi[a+b*x]^2,x]] /;
FreeQ[{a,b},x] && IntegerQ[m] && m<-2 *)
```

$$\int e^{a+b\,x}\,\text{ExpIntegralEi}[c+d\,x]\,dx$$

- **Derivation: Integration by parts**
- **Rule:**

$$\int e^{a+b\,x}\,\text{ExpIntegralEi}[c+d\,x]\,dx \longrightarrow \frac{e^{a+b\,x}\,\text{ExpIntegralEi}[c+d\,x]}{b}-\frac{d}{b}\int\frac{e^{a+b\,x}\,e^{c+d\,x}}{c+d\,x}\,dx$$

- **Program code:**

```
Int[E^(a_.+b_.*x_)*ExpIntegralEi[c_.+d_.*x_],x_Symbol] :=
  E^(a+b*x)*ExpIntegralEi[c+d*x]/b -
  Dist[d/b,Int[E^(a+b*x)*E^(c+d*x)/(c+d*x),x]] /;
FreeQ[{a,b,c,d},x]
```

$$\int x^m e^{a+b x} \text{ExpIntegralEi}[c+d x] \, dx$$

- **Derivation: Integration by parts**
- **Rule: If $m \in \mathbb{Z} \land m > 0$, then**

$$\int x^m e^{a+b x} \text{ExpIntegralEi}[c+d x] \, dx \longrightarrow \frac{x^m e^{a+b x} \text{ExpIntegralEi}[c+d x]}{b} - \frac{d}{b} \int \frac{x^m e^{a+b x} e^{c+d x}}{c+d x} \, dx - \frac{m}{b} \int x^{m-1} e^{a+b x} \text{ExpIntegralEi}[c+d x] \, dx$$

- **Program code:**

```
Int[x_^m_.*E^(a_.+b_.*x_)*ExpIntegralEi[c_.+d_.*x_],x_Symbol] :=
  x^m*E^(a+b*x)*ExpIntegralEi[c+d*x]/b -
  Dist[d/b,Int[x^m*E^(a+b*x)*E^(c+d*x)/(c+d*x),x]] -
  Dist[m/b,Int[x^(m-1)*E^(a+b*x)*ExpIntegralEi[c+d*x],x]] /;
FreeQ[{a,b,c,d},x] && IntegerQ[m] && m>0
```

- **Derivation: Inverted integration by parts**
- **Rule: If $m \in \mathbb{Z} \land m < -1$, then**

$$\int x^m e^{a+b x} \text{ExpIntegralEi}[c+d x] \, dx \longrightarrow \frac{x^{m+1} e^{a+b x} \text{ExpIntegralEi}[c+d x]}{m+1} - \frac{d}{m+1} \int \frac{x^{m+1} e^{a+b x} e^{c+d x}}{c+d x} \, dx - \frac{b}{m+1} \int x^{m+1} e^{a+b x} \text{ExpIntegralEi}[c+d x] \, dx$$

- **Program code:**

```
Int[x_^m_*E^(a_.+b_.*x_)*ExpIntegralEi[c_.+d_.*x_],x_Symbol] :=
  x^(m+1)*E^(a+b*x)*ExpIntegralEi[c+d*x]/(m+1) -
  Dist[d/(m+1),Int[x^(m+1)*E^(a+b*x)*E^(c+d*x)/(c+d*x),x]] -
  Dist[b/(m+1),Int[x^(m+1)*E^(a+b*x)*ExpIntegralEi[c+d*x],x]] /;
FreeQ[{a,b,c,d},x] && IntegerQ[m] && m<-1
```