$$\int \text{SinIntegral}[a+b\,x]^n\,dx$$

- **Derivation: Integration by parts**
- **Rule:**

$$\int \text{SinIntegral}[a+b\,x]\,dx \longrightarrow \frac{(a+b\,x)\ \text{SinIntegral}[a+b\,x]}{b}+\frac{\text{Cos}[a+b\,x]}{b}$$

- **Program code:**

```
Int[SinIntegral[a_.+b_.*x_],x_Symbol] :=
  (a+b*x)*SinIntegral[a+b*x]/b + Cos[a+b*x]/b/;
FreeQ[{a,b},x]
```

```
Int[CosIntegral[a_.+b_.*x_],x_Symbol] :=
  (a+b*x)*CosIntegral[a+b*x]/b - Sin[a+b*x]/b /;
FreeQ[{a,b},x]
```

- **Derivation: Integration by parts**
- **Rule:**

$$\int \text{SinIntegral}[a+b\,x]^2\,dx \longrightarrow \frac{(a+b\,x)\ \text{SinIntegral}[a+b\,x]^2}{b}-2\int \text{Sin}[a+b\,x]\ \text{SinIntegral}[a+b\,x]\,dx$$

- **Program code:**

```
Int[SinIntegral[a_.+b_.*x_]^2,x_Symbol] :=
  (a+b*x)*SinIntegral[a+b*x]^2/b -
  Dist[2,Int[Sin[a+b*x]*SinIntegral[a+b*x],x]] /;
FreeQ[{a,b},x]
```

```
Int[CosIntegral[a_.+b_.*x_]^2,x_Symbol] :=
  (a+b*x)*CosIntegral[a+b*x]^2/b -
  Dist[2,Int[Cos[a+b*x]*CosIntegral[a+b*x],x]] /;
FreeQ[{a,b},x]
```

# $\int x^m \,\text{SinIntegral}[a+b\,x]^n \,dx$

- **Derivation: Integration by parts**
- **Rule: If $m+1 \neq 0$, then**

$$\int x^m \,\text{SinIntegral}[a+b\,x]\,dx \longrightarrow \frac{x^{m+1}\,\text{SinIntegral}[a+b\,x]}{m+1} - \frac{b}{m+1}\int \frac{x^{m+1}\,\text{Sin}[a+b\,x]}{a+b\,x}\,dx$$

- **Program code:**

```
Int[x_^m_.*SinIntegral[a_.+b_.*x_],x_Symbol] :=
  x^(m+1)*SinIntegral[a+b*x]/(m+1) -
  Dist[b/(m+1),Int[x^(m+1)*Sin[a+b*x]/(a+b*x),x]] /;
FreeQ[{a,b,m},x] && NonzeroQ[m+1]
```

```
Int[x_^m_.*CosIntegral[a_.+b_.*x_],x_Symbol] :=
  x^(m+1)*CosIntegral[a+b*x]/(m+1) -
  Dist[b/(m+1),Int[x^(m+1)*Cos[a+b*x]/(a+b*x),x]] /;
FreeQ[{a,b,m},x] && NonzeroQ[m+1]
```

- **Derivation: Integration by parts**
- **Rule: If $m \in \mathbb{Z} \land m > 0$, then**

$$\int x^m \,\text{SinIntegral}[b\,x]^2\,dx \longrightarrow \frac{x^{m+1}\,\text{SinIntegral}[b\,x]^2}{m+1} - \frac{2}{m+1}\int x^m\,\text{Sin}[b\,x]\,\text{SinIntegral}[b\,x]\,dx$$

- **Program code:**

```
Int[x_^m_.*SinIntegral[b_.*x_]^2,x_Symbol] :=
  x^(m+1)*SinIntegral[b*x]^2/(m+1) -
  Dist[2/(m+1),Int[x^m*Sin[b*x]*SinIntegral[b*x],x]] /;
FreeQ[b,x] && IntegerQ[m] && m>0
```

```
Int[x_^m_.*CosIntegral[b_.*x_]^2,x_Symbol] :=
  x^(m+1)*CosIntegral[b*x]^2/(m+1) -
  Dist[2/(m+1),Int[x^m*Cos[b*x]*CosIntegral[b*x],x]] /;
FreeQ[b,x] && IntegerQ[m] && m>0
```

- **Derivation: Iterated integration by parts**
- **Rule: If $m \in \mathbb{Z} \land m > 0$, then**

$$\int x^m \text{SinIntegral}[a+b\,x]^2\,dx \longrightarrow \frac{x^{m+1}\,\text{SinIntegral}[a+b\,x]^2}{m+1}+\frac{a\,x^m\,\text{SinIntegral}[a+b\,x]^2}{b\,(m+1)}-\frac{2}{m+1}\int x^m\,\text{Sin}[a+b\,x]\,\text{SinIntegral}[a+b\,x]\,dx-\frac{a\,m}{b\,(m+1)}\int x^{m-1}\,\text{SinIntegral}[a+b\,x]^2\,dx$$

- **Program code:**

```
Int[x_^m_.*SinIntegral[a_+b_.*x_]^2,x_Symbol] :=
  x^(m+1)*SinIntegral[a+b*x]^2/(m+1) +
  a*x^m*SinIntegral[a+b*x]^2/(b*(m+1)) -
  Dist[2/(m+1),Int[x^m*Sin[a+b*x]*SinIntegral[a+b*x],x]] -
  Dist[a*m/(b*(m+1)),Int[x^(m-1)*SinIntegral[a+b*x]^2,x]] /;
FreeQ[{a,b},x] && IntegerQ[m] && m>0
```

```
Int[x_^m_.*CosIntegral[a_+b_.*x_]^2,x_Symbol] :=
  x^(m+1)*CosIntegral[a+b*x]^2/(m+1) +
  a*x^m*CosIntegral[a+b*x]^2/(b*(m+1)) -
  Dist[2/(m+1),Int[x^m*Cos[a+b*x]*CosIntegral[a+b*x],x]] -
  Dist[a*m/(b*(m+1)),Int[x^(m-1)*CosIntegral[a+b*x]^2,x]] /;
FreeQ[{a,b},x] && IntegerQ[m] && m>0
```

- **Derivation: Inverted integration by parts**
- **Rule: If $m \in \mathbb{Z} \land m < -2$, then**

$$\int x^m \text{SinIntegral}[a+b\,x]^2\,dx \longrightarrow \frac{b\,x^{m+2}\,\text{SinIntegral}[a+b\,x]^2}{a\,(m+1)}+\frac{x^{m+1}\,\text{SinIntegral}[a+b\,x]^2}{m+1}-\frac{2\,b}{a\,(m+1)}\int x^{m+1}\,\text{Sin}[a+b\,x]\,\text{SinIntegral}[a+b\,x]\,dx-\frac{b\,(m+2)}{a\,(m+1)}\int x^{m+1}\,\text{SinIntegral}[a+b\,x]^2\,dx$$

- **Program code:**

```
(* Int[x_^m_.*SinIntegral[a_+b_.*x_]^2,x_Symbol] :=
  b*x^(m+2)*SinIntegral[a+b*x]^2/(a*(m+1)) +
  x^(m+1)*SinIntegral[a+b*x]^2/(m+1) -
  Dist[2*b/(a*(m+1)),Int[x^(m+1)*Sin[a+b*x]*SinIntegral[a+b*x],x]] -
  Dist[b*(m+2)/(a*(m+1)),Int[x^(m+1)*SinIntegral[a+b*x]^2,x]] /;
FreeQ[{a,b},x] && IntegerQ[m] && m<-2 *)
```

```
(* Int[x_^m_.*CosIntegral[a_+b_.*x_]^2,x_Symbol] :=
  b*x^(m+2)*CosIntegral[a+b*x]^2/(a*(m+1)) +
  x^(m+1)*CosIntegral[a+b*x]^2/(m+1) -
  Dist[2*b/(a*(m+1)),Int[x^(m+1)*Cos[a+b*x]*CosIntegral[a+b*x],x]] -
  Dist[b*(m+2)/(a*(m+1)),Int[x^(m+1)*CosIntegral[a+b*x]^2,x]] /;
FreeQ[{a,b},x] && IntegerQ[m] && m<-2 *)
```

$$\int \text{Sin}[a + b\,x]\ \text{SinIntegral}[c + d\,x]\,dx$$

- **Reference: G&R 5.32.2**
- **Derivation: Integration by parts**
- **Rule:**

$$\int \text{Sin}[a + b\,x]\ \text{SinIntegral}[c + d\,x]\,dx \longrightarrow -\frac{\text{Cos}[a + b\,x]\ \text{SinIntegral}[c + d\,x]}{b} + \frac{d}{b}\int \frac{\text{Cos}[a + b\,x]\ \text{Sin}[c + d\,x]}{c + d\,x}\,dx$$

- **Program code:**

```
Int[Sin[a_.+b_.*x_]*SinIntegral[c_.+d_.*x_],x_Symbol] :=
  -Cos[a+b*x]*SinIntegral[c+d*x]/b +
  Dist[d/b,Int[Cos[a+b*x]*Sin[c+d*x]/(c+d*x),x]] /;
FreeQ[{a,b,c,d},x]
```

- **Reference: G&R 5.31.1**

```
Int[Cos[a_.+b_.*x_]*CosIntegral[c_.+d_.*x_],x_Symbol] :=
  Sin[a+b*x]*CosIntegral[c+d*x]/b -
  Dist[d/b,Int[Sin[a+b*x]*Cos[c+d*x]/(c+d*x),x]] /;
FreeQ[{a,b,c,d},x]
```

# $\int x^m \operatorname{Sin}[a+b\,x]\,\operatorname{SinIntegral}[c+d\,x]\,dx$

- **Derivation: Integration by parts**
- **Rule: If $m \in \mathbb{Z} \wedge m > 0$, then**

$$\int x^m \operatorname{Sin}[a+b\,x]\,\operatorname{SinIntegral}[c+d\,x]\,dx \longrightarrow -\frac{x^m \operatorname{Cos}[a+b\,x]\,\operatorname{SinIntegral}[c+d\,x]}{b} + \frac{d}{b}\int \frac{x^m \operatorname{Cos}[a+b\,x]\,\operatorname{Sin}[c+d\,x]}{c+d\,x}\,dx + \frac{m}{b}\int x^{m-1} \operatorname{Cos}[a+b\,x]\,\operatorname{SinIntegral}[c+d\,x]\,dx$$

- **Program code:**

```
Int[x_^m_.*Sin[a_.+b_.*x_]*SinIntegral[c_.+d_.*x_],x_Symbol] :=
  -x^m*Cos[a+b*x]*SinIntegral[c+d*x]/b +
  Dist[d/b,Int[x^m*Cos[a+b*x]*Sin[c+d*x]/(c+d*x),x]] +
  Dist[m/b,Int[x^(m-1)*Cos[a+b*x]*SinIntegral[c+d*x],x]] /;
FreeQ[{a,b,c,d},x] && IntegerQ[m] && m>0
```

```
Int[x_^m_.*Cos[a_.+b_.*x_]*CosIntegral[c_.+d_.*x_],x_Symbol] :=
  x^m*Sin[a+b*x]*CosIntegral[c+d*x]/b -
  Dist[d/b,Int[x^m*Sin[a+b*x]*Cos[c+d*x]/(c+d*x),x]] -
  Dist[m/b,Int[x^(m-1)*Sin[a+b*x]*CosIntegral[c+d*x],x]] /;
FreeQ[{a,b,c,d},x] && IntegerQ[m] && m>0
```

- **Derivation: Inverted integration by parts**
- **Rule: If $m \in \mathbb{Z} \wedge m < -1$, then**

$$\int x^m \operatorname{Sin}[a+b\,x]\,\operatorname{SinIntegral}[c+d\,x]\,dx \longrightarrow \frac{x^{m+1} \operatorname{Sin}[a+b\,x]\,\operatorname{SinIntegral}[c+d\,x]}{m+1} - \frac{d}{m+1}\int \frac{x^{m+1} \operatorname{Sin}[a+b\,x]\,\operatorname{Sin}[c+d\,x]}{c+d\,x}\,dx - \frac{b}{m+1}\int x^{m+1} \operatorname{Cos}[a+b\,x]\,\operatorname{SinIntegral}[c+d\,x]\,dx$$

- **Program code:**

```
Int[x_^m_*Sin[a_.+b_.*x_]*SinIntegral[c_.+d_.*x_],x_Symbol] :=
  x^(m+1)*Sin[a+b*x]*SinIntegral[c+d*x]/(m+1) -
  Dist[d/(m+1),Int[x^(m+1)*Sin[a+b*x]*Sin[c+d*x]/(c+d*x),x]] -
  Dist[b/(m+1),Int[x^(m+1)*Cos[a+b*x]*SinIntegral[c+d*x],x]] /;
FreeQ[{a,b,c,d},x] && IntegerQ[m] && m<-1
```

```
Int[x_^m_.*Cos[a_.+b_.*x_]*CosIntegral[c_.+d_.*x_],x_Symbol] :=
  x^(m+1)*Cos[a+b*x]*CosIntegral[c+d*x]/(m+1) -
  Dist[d/(m+1),Int[x^(m+1)*Cos[a+b*x]*Cos[c+d*x]/(c+d*x),x]] +
  Dist[b/(m+1),Int[x^(m+1)*Sin[a+b*x]*CosIntegral[c+d*x],x]] /;
FreeQ[{a,b,c,d},x] && IntegerQ[m] && m<-1
```

$$\int \mathrm{Cos}[a+b\,x]\ \mathrm{SinIntegral}[c+d\,x]\,dx$$

- **Reference: G&R 5.32.1**
- **Derivation: Integration by parts**
- **Rule:**

$$\int \mathrm{Cos}[a+b\,x]\ \mathrm{SinIntegral}[c+d\,x]\,dx \longrightarrow \frac{\mathrm{Sin}[a+b\,x]\ \mathrm{SinIntegral}[c+d\,x]}{b}-\frac{d}{b}\int\frac{\mathrm{Sin}[a+b\,x]\ \mathrm{Sin}[c+d\,x]}{c+d\,x}\,dx$$

- **Program code:**

```
Int[Cos[a_.+b_.*x_]*SinIntegral[c_.+d_.*x_],x_Symbol] :=
  Sin[a+b*x]*SinIntegral[c+d*x]/b -
  Dist[d/b,Int[Sin[a+b*x]*Sin[c+d*x]/(c+d*x),x]] /;
FreeQ[{a,b,c,d},x]
```

- **Reference: G&R 5.31.2**

```
Int[Sin[a_.+b_.*x_]*CosIntegral[c_.+d_.*x_],x_Symbol] :=
  -Cos[a+b*x]*CosIntegral[c+d*x]/b +
  Dist[d/b,Int[Cos[a+b*x]*Cos[c+d*x]/(c+d*x),x]] /;
FreeQ[{a,b,c,d},x]
```

# $\int x^m \operatorname{Cos}[a+b\,x]\, \operatorname{SinIntegral}[c+d\,x]\, dx$

- **Derivation: Integration by parts**
- **Rule: If $m \in \mathbb{Z} \land m > 0$, then**

$$\int x^m \operatorname{Cos}[a+b\,x]\, \operatorname{SinIntegral}[c+d\,x]\, dx \longrightarrow \frac{x^m \operatorname{Sin}[a+b\,x]\, \operatorname{SinIntegral}[c+d\,x]}{b} - \frac{d}{b} \int \frac{x^m \operatorname{Sin}[a+b\,x]\, \operatorname{Sin}[c+d\,x]}{c+d\,x}\, dx - \frac{m}{b} \int x^{m-1} \operatorname{Sin}[a+b\,x]\, \operatorname{SinIntegral}[c+d\,x]\, dx$$

- **Program code:**

```
Int[x_^m_.*Cos[a_.+b_.*x_]*SinIntegral[c_.+d_.*x_],x_Symbol] :=
  x^m*Sin[a+b*x]*SinIntegral[c+d*x]/b -
  Dist[d/b,Int[x^m*Sin[a+b*x]*Sin[c+d*x]/(c+d*x),x]] -
  Dist[m/b,Int[x^(m-1)*Sin[a+b*x]*SinIntegral[c+d*x],x]] /;
FreeQ[{a,b,c,d},x] && IntegerQ[m] && m>0
```

```
Int[x_^m_.*Sin[a_.+b_.*x_]*CosIntegral[c_.+d_.*x_],x_Symbol] :=
  -x^m*Cos[a+b*x]*CosIntegral[c+d*x]/b +
  Dist[d/b,Int[x^m*Cos[a+b*x]*Cos[c+d*x]/(c+d*x),x]] +
  Dist[m/b,Int[x^(m-1)*Cos[a+b*x]*CosIntegral[c+d*x],x]] /;
FreeQ[{a,b,c,d},x] && IntegerQ[m] && m>0
```

- **Derivation: Inverted integration by parts**
- **Rule: If $m \in \mathbb{Z} \land m < -1$, then**

$$\int x^m \operatorname{Cos}[a+b\,x]\, \operatorname{SinIntegral}[c+d\,x]\, dx \longrightarrow \frac{x^{m+1} \operatorname{Cos}[a+b\,x]\, \operatorname{SinIntegral}[c+d\,x]}{m+1} - \frac{d}{m+1} \int \frac{x^{m+1} \operatorname{Cos}[a+b\,x]\, \operatorname{Sin}[c+d\,x]}{c+d\,x}\, dx + \frac{b}{m+1} \int x^{m+1} \operatorname{Sin}[a+b\,x]\, \operatorname{SinIntegral}[c+d\,x]\, dx$$

- **Program code:**

```
Int[x_^m_.*Cos[a_.+b_.*x_]*SinIntegral[c_.+d_.*x_],x_Symbol] :=
  x^(m+1)*Cos[a+b*x]*SinIntegral[c+d*x]/(m+1) -
  Dist[d/(m+1),Int[x^(m+1)*Cos[a+b*x]*Sin[c+d*x]/(c+d*x),x]] +
  Dist[b/(m+1),Int[x^(m+1)*Sin[a+b*x]*SinIntegral[c+d*x],x]] /;
FreeQ[{a,b,c,d},x] && IntegerQ[m] && m<-1
```

```
Int[x_^m_*Sin[a_.+b_.*x_]*CosIntegral[c_.+d_.*x_],x_Symbol] :=
  x^(m+1)*Sin[a+b*x]*CosIntegral[c+d*x]/(m+1) -
  Dist[d/(m+1),Int[x^(m+1)*Sin[a+b*x]*Cos[c+d*x]/(c+d*x),x]] -
  Dist[b/(m+1),Int[x^(m+1)*Cos[a+b*x]*CosIntegral[c+d*x],x]] /;
FreeQ[{a,b,c,d},x] && IntegerQ[m] && m<-1
```