$$\int \text{ArcTan}[a\,x]^n \, dx$$

- **Reference: G&R 2.822.1, CRC 443, A&S 4.4.60**
- **Derivation: Integration by parts**
- **Rule:**

$$\int \text{ArcTan}[a\,x] \, dx \longrightarrow x\,\text{ArcTan}[a\,x] - \frac{\text{Log}\left[1+a^2\,x^2\right]}{2\,a}$$

- **Program code:**

```
Int[ArcTan[a_.*x_],x_Symbol] :=
  x*ArcTan[a*x] - Log[1+a^2*x^2]/(2*a) /;
FreeQ[a,x]
```

- **Derivation: Integration by parts**
- **Rule: If** $n \in \mathbb{Z} \wedge n > 1$**, then**

$$\int \text{ArcTan}[a\,x]^n \, dx \longrightarrow x\,\text{ArcTan}[a\,x]^n - a\,n \int \frac{x\,\text{ArcTan}[a\,x]^{n-1}}{1+a^2\,x^2} \, dx$$

- **Program code:**

```
Int[ArcTan[a_.*x_]^n_,x_Symbol] :=
  x*ArcTan[a*x]^n -
  Dist[a*n,Int[x*ArcTan[a*x]^(n-1)/(1+a^2*x^2),x]] /;
FreeQ[a,x] && IntegerQ[n] && n>1
```

$$\int x^m \text{ArcTan}[a\,x]^n \, dx$$

- **Derivation: Iterated integration by parts**
- **Rule: If $n \in \mathbb{Z} \land n > 0$, then**

$$\int x \text{ ArcTan}[a\,x]^n \, dx \longrightarrow \frac{\text{ArcTan}[a\,x]^n}{2\,a^2} + \frac{x^2 \text{ ArcTan}[a\,x]^n}{2} - \frac{n}{2\,a} \int \text{ArcTan}[a\,x]^{n-1} \, dx$$

- **Program code:**

```
Int[x_*ArcTan[a_.*x_]^n_.,x_Symbol] :=
  ArcTan[a*x]^n/(2*a^2) + x^2*ArcTan[a*x]^n/2 -
  Dist[n/(2*a),Int[ArcTan[a*x]^(n-1),x]] /;
FreeQ[a,x] && IntegerQ[n] && n>0
```

- **Derivation: Iterated integration by parts**
- **Rule: If $n \in \mathbb{Z} \land n > 0 \land m > 1$, then**

$$\int x^m \text{ ArcTan}[a\,x]^n \, dx \longrightarrow \frac{x^{m-1} \text{ ArcTan}[a\,x]^n}{a^2\,(m+1)} + \frac{x^{m+1} \text{ ArcTan}[a\,x]^n}{m+1} - \frac{n}{a\,(m+1)} \int x^{m-1} \text{ ArcTan}[a\,x]^{n-1} \, dx - \frac{m-1}{a^2\,(m+1)} \int x^{m-2} \text{ ArcTan}[a\,x]^n \, dx$$

- **Program code:**

```
Int[x_^m_*ArcTan[a_.*x_]^n_.,x_Symbol] :=
  x^(m-1)*ArcTan[a*x]^n/(a^2*(m+1)) + x^(m+1)*ArcTan[a*x]^n/(m+1) -
  Dist[n/(a*(m+1)),Int[x^(m-1)*ArcTan[a*x]^(n-1),x]] -
  Dist[(m-1)/(a^2*(m+1)),Int[x^(m-2)*ArcTan[a*x]^n,x]] /;
FreeQ[a,x] && IntegerQ[n] && n>0 && RationalQ[m] && m>1
```

- **Derivation: Integration by parts**
- **Rule: If $n \in \mathbb{Z} \wedge n > 1$, then**

$$\int \frac{\text{ArcTan}[a\,x]^n}{x}\,dx \longrightarrow 2\,\text{ArcTan}[a\,x]^n\,\text{ArcTanh}\left[1-\frac{2\,I}{I-a\,x}\right]-2\,a\,n\int \frac{\text{ArcTan}[a\,x]^{n-1}\,\text{ArcTanh}\left[1-\frac{2\,I}{I-a\,x}\right]}{1+a^2\,x^2}\,dx$$

- **Program code:**

```
Int[ArcTan[a_.*x_]^n_/x_,x_Symbol] :=
  2*ArcTan[a*x]^n*ArcTanh[1-2*I/(I-a*x)] -
  Dist[2*a*n,Int[ArcTan[a*x]^(n-1)*ArcTanh[1-2*I/(I-a*x)]/(1+a^2*x^2),x]] /;
FreeQ[a,x] && IntegerQ[n] && n>1
```

- **Derivation: Integration by parts**
- **Rule: If $n \in \mathbb{Z} \wedge n > 0$, then**

$$\int \frac{\text{ArcTan}[a\,x]^n}{x^2}\,dx \longrightarrow -\frac{\text{ArcTan}[a\,x]^n}{x}+a\,n\int \frac{\text{ArcTan}[a\,x]^{n-1}}{x\left(1+a^2\,x^2\right)}\,dx$$

- **Program code:**

```
Int[ArcTan[a_.*x_]^n_./x_^2,x_Symbol] :=
  -ArcTan[a*x]^n/x +
  Dist[a*n,Int[ArcTan[a*x]^(n-1)/(x*(1+a^2*x^2)),x]] /;
FreeQ[a,x] && IntegerQ[n] && n>0
```

- **Derivation: Inverted iterated integration by parts special case**
- **Rule: If $n \in \mathbb{Z} \wedge n > 0$, then**

$$\int \frac{\text{ArcTan}[a\,x]^n}{x^3}\,dx \longrightarrow -\frac{a^2\,\text{ArcTan}[a\,x]^n}{2}-\frac{\text{ArcTan}[a\,x]^n}{2\,x^2}+\frac{a\,n}{2}\int \frac{\text{ArcTan}[a\,x]^{n-1}}{x^2}\,dx$$

- **Program code:**

```
Int[ArcTan[a_.*x_]^n_./x_^3,x_Symbol] :=
  -a^2*ArcTan[a*x]^n/2 - ArcTan[a*x]^n/(2*x^2) +
  Dist[a*n/2,Int[ArcTan[a*x]^(n-1)/x^2,x]] /;
FreeQ[a,x] && IntegerQ[n] && n>0
```

- **Derivation: Inverted iterated integration by parts**
- **Rule: If $n \in \mathbb{Z} \land n > 0 \land m < -3$, then**

$$\int x^m \operatorname{ArcTan}[a\,x]^n\,dx \longrightarrow \frac{x^{m+1}\operatorname{ArcTan}[a\,x]^n}{m+1}+\frac{a^2\,x^{m+3}\operatorname{ArcTan}[a\,x]^n}{m+1}-\frac{a\,n}{m+1}\int x^{m+1}\operatorname{ArcTan}[a\,x]^{n-1}\,dx-\frac{a^2\,(m+3)}{m+1}\int x^{m+2}\operatorname{ArcTan}[a\,x]^n\,dx$$

- **Program code:**

```
Int[x_^m_*ArcTan[a_.*x_]^n_.,x_Symbol] :=
  x^(m+1)*ArcTan[a*x]^n/(m+1) + a^2*x^(m+3)*ArcTan[a*x]^n/(m+1) -
  Dist[a*n/(m+1),Int[x^(m+1)*ArcTan[a*x]^(n-1),x]] -
  Dist[a^2*(m+3)/(m+1),Int[x^(m+2)*ArcTan[a*x]^n,x]] /;
FreeQ[a,x] && IntegerQ[n] && n>0 && RationalQ[m] && m<-3
```

$$\int \frac{\text{ArcTan}[a\,x]^n}{c+d\,x}\,dx$$

- **Derivation: Integration by parts**

- **Rule: If** $a^2 c^2 + d^2 = 0 \wedge n \in \mathbb{Z} \wedge n > 0$**, then**

$$\int \frac{\text{ArcTan}[a\,x]^n}{c+d\,x}\,dx \longrightarrow -\frac{\text{ArcTan}[a\,x]^n\,\text{Log}\left[\frac{2\,c}{c+d\,x}\right]}{d} + \frac{a\,n}{d}\int \frac{\text{ArcTan}[a\,x]^{n-1}\,\text{Log}\left[\frac{2\,c}{c+d\,x}\right]}{1+a^2\,x^2}\,dx$$

- **Program code:**

```
Int[ArcTan[a_.*x_]^n_./(c_+d_.*x_),x_Symbol] :=
  -ArcTan[a*x]^n*Log[2*c/(c+d*x)]/d +
  Dist[a*n/d,Int[ArcTan[a*x]^(n-1)*Log[2*c/(c+d*x)]/(1+a^2*x^2),x]] /;
FreeQ[{a,c,d},x] && ZeroQ[a^2*c^2+d^2] && IntegerQ[n] && n>0
```

$$\int \frac{x^m \operatorname{ArcTan}[a\,x]^n}{c+d\,x}\,dx$$

- **Derivation: Integration by parts**

- **Rule: If** $a^2 c^2 + d^2 = 0 \wedge n \in \mathbb{Z} \wedge n > 0$, **then**

$$\int \frac{\operatorname{ArcTan}[a\,x]^n}{x\,(c+d\,x)}\,dx \longrightarrow \frac{\operatorname{ArcTan}[a\,x]^n \operatorname{Log}\left[2-\frac{2\,c}{c+d\,x}\right]}{c} - \frac{a\,n}{c}\int \frac{\operatorname{ArcTan}[a\,x]^{n-1} \operatorname{Log}\left[2-\frac{2\,c}{c+d\,x}\right]}{1+a^2\,x^2}\,dx$$

- **Program code:**

```
Int[ArcTan[a_.*x_]^n_./(x_*(c_+d_.*x_)),x_Symbol] :=
  ArcTan[a*x]^n*Log[2-2*c/(c+d*x)]/c -
  Dist[a*n/c,Int[ArcTan[a*x]^(n-1)*Log[2-2*c/(c+d*x)]/(1+a^2*x^2),x]] /;
FreeQ[{a,c,d},x] && ZeroQ[a^2*c^2+d^2] && IntegerQ[n] && n>0
```

- **Derivation: Integration by parts**

- **Rule: If** $a^2 c^2 + d^2 = 0 \wedge n \in \mathbb{Z} \wedge n > 0$, **then**

$$\int \frac{\operatorname{ArcTan}[a\,x]^n}{c\,x+d\,x^2}\,dx \longrightarrow \frac{\operatorname{ArcTan}[a\,x]^n \operatorname{Log}\left[2-\frac{2\,c}{c+d\,x}\right]}{c} - \frac{a\,n}{c}\int \frac{\operatorname{ArcTan}[a\,x]^{n-1} \operatorname{Log}\left[2-\frac{2\,c}{c+d\,x}\right]}{1+a^2\,x^2}\,dx$$

- **Program code:**

```
Int[ArcTan[a_.*x_]^n_./(c_.*x_+d_.*x_^2),x_Symbol] :=
  ArcTan[a*x]^n*Log[2-2*c/(c+d*x)]/c -
  Dist[a*n/c,Int[ArcTan[a*x]^(n-1)*Log[2-2*c/(c+d*x)]/(1+a^2*x^2),x]] /;
FreeQ[{a,c,d},x] && ZeroQ[a^2*c^2+d^2] && IntegerQ[n] && n>0
```

- **Derivation: Algebraic simplification**
- **Basis:** $\frac{x}{c+d\,x} == \frac{1}{d} - \frac{c}{d\,(c+d\,x)}$
- **Rule: If** $a^2\,c^2 + d^2 = 0 \wedge m > 0 \wedge n \in \mathbb{Z} \wedge n > 0$**, then**

$$\int \frac{x^m\,\text{ArcTan}[a\,x]^n}{c+d\,x}\,dx \longrightarrow \frac{1}{d}\int x^{m-1}\,\text{ArcTan}[a\,x]^n\,dx - \frac{c}{d}\int \frac{x^{m-1}\,\text{ArcTan}[a\,x]^n}{c+d\,x}\,dx$$

- **Program code:**

```
Int[x_^m_.*ArcTan[a_.*x_]^n_./(c_+d_.*x_),x_Symbol] :=
  Dist[1/d,Int[x^(m-1)*ArcTan[a*x]^n,x]] -
  Dist[c/d,Int[x^(m-1)*ArcTan[a*x]^n/(c+d*x),x]] /;
FreeQ[{a,c,d},x] && ZeroQ[a^2*c^2+d^2] && RationalQ[m] && m>0 && IntegerQ[n] && n>0
```

- **Derivation: Algebraic simplification**
- **Basis:** $\frac{1}{c+d\,x} == \frac{1}{c} - \frac{d\,x}{c\,(c+d\,x)}$
- **Rule: If** $a^2\,c^2 + d^2 = 0 \wedge m < -1 \wedge n \in \mathbb{Z} \wedge n > 0$**, then**

$$\int \frac{x^m\,\text{ArcTan}[a\,x]^n}{c+d\,x}\,dx \longrightarrow \frac{1}{c}\int x^{m}\,\text{ArcTan}[a\,x]^n\,dx - \frac{d}{c}\int \frac{x^{m+1}\,\text{ArcTan}[a\,x]^n}{c+d\,x}\,dx$$

- **Program code:**

```
Int[x_^m_*ArcTan[a_.*x_]^n_./(c_+d_.*x_),x_Symbol] :=
  Dist[1/c,Int[x^m*ArcTan[a*x]^n,x]] -
  Dist[d/c,Int[x^(m+1)*ArcTan[a*x]^n/(c+d*x),x]] /;
FreeQ[{a,c,d},x] && ZeroQ[a^2*c^2+d^2] && RationalQ[m] && m<-1 && IntegerQ[n] && n>0
```

$$\int \frac{\text{ArcTan}[a\,x]^n}{c + d\,x^2}\,dx$$

- **Derivation: Reciprocal rule for integration**
- **Rule: If $d = a^2 c$, then**

$$\int \frac{1}{\left(c + d\,x^2\right)\,\text{ArcTan}[a\,x]}\,dx \longrightarrow \frac{\text{Log}[\text{ArcTan}[a\,x]]}{a\,c}$$

- **Program code:**

```
Int[1/((c_+d_.*x_^2)*ArcTan[a_.*x_]),x_Symbol] :=
  Log[ArcTan[a*x]]/(a*c) /;
FreeQ[{a,c,d},x] && ZeroQ[d-a^2*c]
```

- **Derivation: Power rule for integration**
- **Rule: If $d = a^2 c \wedge n \neq -1$, then**

$$\int \frac{\text{ArcTan}[a\,x]^n}{c + d\,x^2}\,dx \longrightarrow \frac{\text{ArcTan}[a\,x]^{n+1}}{a\,c\,(n+1)}$$

- **Program code:**

```
Int[ArcTan[a_.*x_]^n_./(c_+d_.*x_^2),x_Symbol] :=
  ArcTan[a*x]^(n+1)/(a*c*(n+1)) /;
FreeQ[{a,c,d,n},x] && ZeroQ[d-a^2*c] && NonzeroQ[n+1]
```

$$\int \frac{x^m \operatorname{ArcTan}[a\,x]^n}{c+d\,x^2}\,dx$$

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{x}{1+a^2x^2} = -\frac{I}{a\,(1+a^2x^2)} - \frac{1}{a\,(I-a\,x)}$
- **Rule: If** $d = a^2 c \wedge n > 0$, **then**

$$\int \frac{x \operatorname{ArcTan}[a\,x]^n}{c+d\,x^2}\,dx \longrightarrow -\frac{I \operatorname{ArcTan}[a\,x]^{n+1}}{d\,(n+1)} - \frac{1}{a\,c}\int \frac{\operatorname{ArcTan}[a\,x]^n}{I-a\,x}\,dx$$

- **Program code:**

```
Int[x_*ArcTan[a_.*x_]^n_./(c_+d_.*x_^2),x_Symbol] :=
  -I*ArcTan[a*x]^(n+1)/(d*(n+1)) -
  Dist[1/(a*c),Int[ArcTan[a*x]^n/(I-a*x),x]] /;
FreeQ[{a,c,d},x] && ZeroQ[d-a^2*c] && RationalQ[n] && n>0
```

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{1}{x\,(1+a^2x^2)} = -\frac{a\,I}{1+a^2x^2} + \frac{I}{x\,(I+a\,x)}$
- **Rule: If** $d = a^2 c \wedge n > 0$, **then**

$$\int \frac{\operatorname{ArcTan}[a\,x]^n}{x\,(c+d\,x^2)}\,dx \longrightarrow -\frac{I \operatorname{ArcTan}[a\,x]^{n+1}}{c\,(n+1)} + \frac{I}{c}\int \frac{\operatorname{ArcTan}[a\,x]^n}{x\,(I+a\,x)}\,dx$$

- **Program code:**

```
Int[ArcTan[a_.*x_]^n_./(x_*(c_+d_.*x_^2)),x_Symbol] :=
  -I*ArcTan[a*x]^(n+1)/(c*(n+1)) +
  Dist[I/c,Int[ArcTan[a*x]^n/(x*(I+a*x)),x]] /;
FreeQ[{a,c,d},x] && ZeroQ[d-a^2*c] && RationalQ[n] && n>0
```

- **Derivation: Algebraic expansion**

- **Basis:** $\frac{1}{x\left(1+a^2 x^2\right)} = -\frac{a\,I}{1+a^2 x^2} + \frac{I}{x\,(I+a\,x)}$

- **Rule: If** $d = a^2 c \wedge n > 0$**, then**

$$\int \frac{\text{ArcTan}[a\,x]^n}{c\,x+d\,x^3}\,dx \longrightarrow -\frac{I\,\text{ArcTan}[a\,x]^{n+1}}{c\,(n+1)} + \frac{I}{c}\int \frac{\text{ArcTan}[a\,x]^n}{x\,(I+a\,x)}\,dx$$

- **Program code:**

```
Int[ArcTan[a_.*x_]^n_./(c_.*x_+d_.*x_^3),x_Symbol] :=
  -I*ArcTan[a*x]^(n+1)/(c*(n+1)) +
  Dist[I/c,Int[ArcTan[a*x]^n/(x*(I+a*x)),x]] /;
FreeQ[{a,c,d},x] && ZeroQ[d-a^2*c] && RationalQ[n] && n>0
```

- **Derivation: Algebraic expansion**

- **Basis:** $\frac{x^2}{c+d\,x^2} = \frac{1}{d} - \frac{c}{d\left(c+d\,x^2\right)}$

- **Rule: If** $d = a^2 c \wedge m > 1 \wedge n > 0$**, then**

$$\int \frac{x^m\,\text{ArcTan}[a\,x]^n}{c+d\,x^2}\,dx \longrightarrow \frac{1}{d}\int x^{m-2}\,\text{ArcTan}[a\,x]^n\,dx - \frac{c}{d}\int \frac{x^{m-2}\,\text{ArcTan}[a\,x]^n}{c+d\,x^2}\,dx$$

- **Program code:**

```
Int[x_^m_*ArcTan[a_.*x_]^n_./(c_+d_.*x_^2),x_Symbol] :=
  Dist[1/d,Int[x^(m-2)*ArcTan[a*x]^n,x]] -
  Dist[c/d,Int[x^(m-2)*ArcTan[a*x]^n/(c+d*x^2),x]] /;
FreeQ[{a,c,d},x] && ZeroQ[d-a^2*c] && RationalQ[{m,n}] && m>1 && n>0
```

- **Derivation: Algebraic expansion**

- **Basis:** $\frac{1}{c+d\,x^2} = \frac{1}{c} - \frac{d\,x^2}{c\left(c+d\,x^2\right)}$

- **Rule: If** $d = a^2 c \wedge m < -1 \wedge n > 0$**, then**

$$\int \frac{x^m\,\text{ArcTan}[a\,x]^n}{c+d\,x^2}\,dx \longrightarrow \frac{1}{c}\int x^m\,\text{ArcTan}[a\,x]^n\,dx - \frac{d}{c}\int \frac{x^{m+2}\,\text{ArcTan}[a\,x]^n}{c+d\,x^2}\,dx$$

- **Program code:**

```
Int[x_^m_*ArcTan[a_.*x_]^n_./(c_+d_.*x_^2),x_Symbol] :=
  Dist[1/c,Int[x^m*ArcTan[a*x]^n,x]] -
  Dist[d/c,Int[x^(m+2)*ArcTan[a*x]^n/(c+d*x^2),x]] /;
FreeQ[{a,c,d},x] && ZeroQ[d-a^2*c] && RationalQ[{m,n}] && m<-1 && n>0
```

- **Derivation: Integration by substitution**

- **Basis: If** $m \in \mathbb{Z}$ **or** $a > 0$, $\frac{x^m \text{ArcTan}[a\,x]^n}{1+a^2\,x^2} = \frac{\text{Tan}[\text{ArcTan}[a\,x]]^m\,\text{ArcTan}[a\,x]^n}{a^{m+1}}\,\partial_x \text{ArcTan}[a\,x]$

- **Rule: If** $d = a^2\,c \,\wedge\, m,\, n \in \mathbb{Q} \,\wedge\, (n < 0 \,\vee\, n \notin \mathbb{Z}) \,\wedge\, (m \in \mathbb{Z} \,\vee\, a > 0)$, **then**

$$\int \frac{x^m \text{ArcTan}[a\,x]^n}{c + d\,x^2}\,dx \;\longrightarrow\; \frac{1}{a^{m+1}\,c}\,\text{Subst}\left[\int x^n\,\text{Tan}[x]^m\,dx,\,x,\,\text{ArcTan}[a\,x]\right]$$

- **Program code:**

```
Int[x_^m_.*ArcTan[a_.*x_]^n_/(c_+d_.*x_^2),x_Symbol] :=
  Dist[1/(a^(m+1)*c),Subst[Int[x^n*Tan[x]^m,x],x,ArcTan[a*x]]] /;
FreeQ[{a,c,d},x] && ZeroQ[d-a^2*c] && RationalQ[{m,n}] && (n<0 || Not[IntegerQ[n]]) && (IntegerQ[m]
```

- **Derivation: Integration by substitution**

- **Basis:** $\frac{x^m \text{ArcTan}[a\,x]^n}{1+a^2\,x^2} = \frac{1}{a}\left(\frac{\text{Tan}[\text{ArcTan}[a\,x]]}{a}\right)^m \text{ArcTan}[a\,x]^n\,\partial_x \text{ArcTan}[a\,x]$

- **Rule: If** $d = a^2\,c \,\wedge\, m,\, n \in \mathbb{Q} \,\wedge\, (n < 0 \,\vee\, n \notin \mathbb{Z}) \,\wedge\, \neg\,(m \in \mathbb{Z} \,\vee\, a > 0)$, **then**

$$\int \frac{x^m \text{ArcTan}[a\,x]^n}{c + d\,x^2}\,dx \;\longrightarrow\; \frac{1}{a\,c}\,\text{Subst}\left[\int x^n \left(\frac{\text{Tan}[x]}{a}\right)^m dx,\,x,\,\text{ArcTan}[a\,x]\right]$$

- **Program code:**

```
Int[x_^m_.*ArcTan[a_.*x_]^n_/(c_+d_.*x_^2),x_Symbol] :=
  Dist[1/(a*c),Subst[Int[x^n*(Tan[x]/a)^m,x],x,ArcTan[a*x]]] /;
FreeQ[{a,c,d},x] && ZeroQ[d-a^2*c] && RationalQ[{m,n}] && (n<0 || Not[IntegerQ[n]]) && Not[IntegerQ[
```

$$\int \frac{\text{ArcTan}[a\,x]^n\,\text{ArcTanh}[u]}{c+d\,x^2}\,dx$$

- **Derivation: Algebraic simplification**
- **Basis:** $\text{ArcTanh}[z] = \frac{1}{2}\,\text{Log}[1+z] - \frac{1}{2}\,\text{Log}[1-z]$
- **Rule: If** $d = a^2\,c \bigwedge n > 0 \bigwedge \left(u^2 = \left(1 - \frac{2\,I}{I+a\,x}\right)^2 \bigvee u^2 = \left(1 - \frac{2\,I}{I-a\,x}\right)^2\right)$**, then**

$$\int \frac{\text{ArcTan}[a\,x]^n\,\text{ArcTanh}[u]}{c+d\,x^2}\,dx \longrightarrow \frac{1}{2}\int \frac{\text{ArcTan}[a\,x]^n\,\text{Log}[1+u]}{c+d\,x^2}\,dx - \frac{1}{2}\int \frac{\text{ArcTan}[a\,x]^n\,\text{Log}[1-u]}{c+d\,x^2}\,dx$$

- **Program code:**

```
Int[ArcTan[a_.*x_]^n_.*ArcTanh[u_]/(c_+d_.*x_^2),x_Symbol] :=
  Dist[1/2,Int[ArcTan[a*x]^n*Log[1+u]/(c+d*x^2),x]] -
  Dist[1/2,Int[ArcTan[a*x]^n*Log[1-u]/(c+d*x^2),x]] /;
FreeQ[{a,c,d},x] && ZeroQ[d-a^2*c] && RationalQ[n] && n>0 && (ZeroQ[u^2-(1-2*I/(I+a*x))^2] || ZeroQ[
```

$$\int \frac{\text{ArcTan}[a\,x]^n\,\text{Log}[u]}{c+d\,x^2}\,dx$$

- **Derivation: Integration by parts**

- **Rule: If** $d=a^2\,c\,\wedge\,n>0\,\wedge\,(1-u)^2=\left(1-\frac{2\,I}{I+a\,x}\right)^2$**, then**

$$\int \frac{\text{ArcTan}[a\,x]^n\,\text{Log}[u]}{c+d\,x^2}\,dx \;\longrightarrow\; \frac{I\,\text{ArcTan}[a\,x]^n\,\text{PolyLog}[2,\,1-u]}{2\,a\,c}-\frac{n\,I}{2}\int \frac{\text{ArcTan}[a\,x]^{n-1}\,\text{PolyLog}[2,\,1-u]}{c+d\,x^2}\,dx$$

- **Program code:**

```
Int[ArcTan[a_.*x_]^n_.*Log[u_]/(c_+d_.*x_^2),x_Symbol] :=
  I*ArcTan[a*x]^n*PolyLog[2,1-u]/(2*a*c) -
  Dist[n*I/2,Int[ArcTan[a*x]^(n-1)*PolyLog[2,1-u]/(c+d*x^2),x]] /;
FreeQ[{a,c,d},x] && ZeroQ[d-a^2*c] && RationalQ[n] && n>0 && ZeroQ[(1-u)^2-(1-2*I/(I+a*x))^2]
```

- **Derivation: Integration by parts**

- **Rule: If** $d=a^2\,c\,\wedge\,n>0\,\wedge\,(1-u)^2=\left(1-\frac{2\,I}{I-a\,x}\right)^2$**, then**

$$\int \frac{\text{ArcTan}[a\,x]^n\,\text{Log}[u]}{c+d\,x^2}\,dx \;\longrightarrow\; -\frac{I\,\text{ArcTan}[a\,x]^n\,\text{PolyLog}[2,\,1-u]}{2\,a\,c}+\frac{n\,I}{2}\int \frac{\text{ArcTan}[a\,x]^{n-1}\,\text{PolyLog}[2,\,1-u]}{c+d\,x^2}\,dx$$

- **Program code:**

```
Int[ArcTan[a_.*x_]^n_.*Log[u_]/(c_+d_.*x_^2),x_Symbol] :=
  -I*ArcTan[a*x]^n*PolyLog[2,1-u]/(2*a*c) +
  Dist[n*I/2,Int[ArcTan[a*x]^(n-1)*PolyLog[2,1-u]/(c+d*x^2),x]] /;
FreeQ[{a,c,d},x] && ZeroQ[d-a^2*c] && RationalQ[n] && n>0 && ZeroQ[(1-u)^2-(1-2*I/(I-a*x))^2]
```

$$\int \frac{\text{ArcTan}[a\,x]^n\,\text{PolyLog}[p,\,u]}{c+d\,x^2}\,dx$$

- **Derivation: Integration by parts**
- **Rule: If** $d=a^2\,c\,\wedge\,n>0\,\wedge\,u^2=\left(1-\frac{2\,I}{I+a\,x}\right)^2$**, then**

$$\int \frac{\text{ArcTan}[a\,x]^n\,\text{PolyLog}[p,\,u]}{c+d\,x^2}\,dx\,\longrightarrow\,-\frac{I\,\text{ArcTan}[a\,x]^n\,\text{PolyLog}[p+1,\,u]}{2\,a\,c}+\frac{n\,I}{2}\int \frac{\text{ArcTan}[a\,x]^{n-1}\,\text{PolyLog}[p+1,\,u]}{c+d\,x^2}\,dx$$

- **Program code:**

```
Int[ArcTan[a_.*x_]^n_.*PolyLog[p_,u_]/(c_+d_.*x_^2),x_Symbol] :=
  -I*ArcTan[a*x]^n*PolyLog[p+1,u]/(2*a*c) +
  Dist[n*I/2,Int[ArcTan[a*x]^(n-1)*PolyLog[p+1,u]/(c+d*x^2),x]] /;
FreeQ[{a,c,d,p},x] && ZeroQ[d-a^2*c] && RationalQ[n] && n>0 && ZeroQ[u^2-(1-2*I/(I+a*x))^2]
```

- **Derivation: Integration by parts**
- **Rule: If** $d=a^2\,c\,\wedge\,n>0\,\wedge\,u^2=\left(1-\frac{2\,I}{I-a\,x}\right)^2$**, then**

$$\int \frac{\text{ArcTan}[a\,x]^n\,\text{PolyLog}[p,\,u]}{c+d\,x^2}\,dx\,\longrightarrow\,\frac{I\,\text{ArcTan}[a\,x]^n\,\text{PolyLog}[p+1,\,u]}{2\,a\,c}-\frac{n\,I}{2}\int \frac{\text{ArcTan}[a\,x]^{n-1}\,\text{PolyLog}[p+1,\,u]}{c+d\,x^2}\,dx$$

- **Program code:**

```
Int[ArcTan[a_.*x_]^n_.*PolyLog[p_,u_]/(c_+d_.*x_^2),x_Symbol] :=
  I*ArcTan[a*x]^n*PolyLog[p+1,u]/(2*a*c) -
  Dist[n*I/2,Int[ArcTan[a*x]^(n-1)*PolyLog[p+1,u]/(c+d*x^2),x]] /;
FreeQ[{a,c,d,p},x] && ZeroQ[d-a^2*c] && RationalQ[n] && n>0 && ZeroQ[u^2-(1-2*I/(I-a*x))^2]
```

$$\int \frac{\text{ArcCot}[a\,x]^m\,\text{ArcTan}[a\,x]^n}{c+d\,x^2}\,dx$$

- **Rule: If** $d = a^2 c$, **then**

$$\int \frac{1}{\left(c+d\,x^2\right)\,\text{ArcCot}[a\,x]\,\text{ArcTan}[a\,x]}\,dx \longrightarrow \frac{-\text{Log}[\text{ArcCot}[a\,x]]+\text{Log}[\text{ArcTan}[a\,x]]}{a\,c\,\text{ArcCot}[a\,x]+a\,c\,\text{ArcTan}[a\,x]}$$

- **Program code:**

```
Int[1/(ArcCot[a_.*x_]*ArcTan[a_.*x_]*(c_+d_.*x_^2)),x_Symbol] :=
  (-Log[ArcCot[a*x]]+Log[ArcTan[a*x]])/(a*c*ArcCot[a*x]+a*c*ArcTan[a*x]) /;
FreeQ[{a,c,d},x] && ZeroQ[d-a^2*c]
```

- **Derivation: Integration by parts**
- **Rule: If** $d = a^2 c \wedge m,\ n \in \mathbb{Z} \wedge 0 < n \le m$, **then**

$$\int \frac{\text{ArcCot}[a\,x]^m\,\text{ArcTan}[a\,x]^n}{c+d\,x^2}\,dx \longrightarrow$$
$$-\frac{\text{ArcCot}[a\,x]^{m+1}\,\text{ArcTan}[a\,x]^n}{a\,c\,(m+1)}+\frac{n}{m+1}\int \frac{\text{ArcCot}[a\,x]^{m+1}\,\text{ArcTan}[a\,x]^{n-1}}{c+d\,x^2}\,dx$$

- **Program code:**

```
Int[ArcCot[a_.*x_]^m_.*ArcTan[a_.*x_]^n_./(c_+d_.*x_^2),x_Symbol] :=
  -ArcCot[a*x]^(m+1)*ArcTan[a*x]^n/(a*c*(m+1)) +
  Dist[n/(m+1),Int[ArcCot[a*x]^(m+1)*ArcTan[a*x]^(n-1)/(c+d*x^2),x]] /;
FreeQ[{a,c,d},x] && ZeroQ[d-a^2*c] && IntegersQ[m,n] && 0<n≤m
```

$$\int \left(c + d\,x^2\right)^m \mathrm{ArcTan}[a\,x]^n\,dx$$

- **Rule: If** $d = a^2 c \wedge c > 0$, **then**

$$\int \frac{\mathrm{ArcTan}[a\,x]}{\sqrt{c + d\,x^2}}\,dx \longrightarrow -\frac{2\,I\,\mathrm{ArcTan}[a\,x]\,\mathrm{ArcTan}\left[\frac{\sqrt{1+I\,a\,x}}{\sqrt{1-I\,a\,x}}\right]}{a\sqrt{c}} + \frac{i\,\mathrm{PolyLog}\left[2, -\frac{i\sqrt{1+I\,a\,x}}{\sqrt{1-I\,a\,x}}\right]}{a\sqrt{c}} - \frac{i\,\mathrm{PolyLog}\left[2, \frac{i\sqrt{1+I\,a\,x}}{\sqrt{1-I\,a\,x}}\right]}{a\sqrt{c}}$$

- **Program code:**

```
Int[ArcTan[a_.*x_]/Sqrt[c_+d_.*x_^2],x_Symbol] :=
  -2*I*ArcTan[a*x]*ArcTan[Sqrt[1+I*a*x]/Sqrt[1-I*a*x]]/(a*Sqrt[c]) +
  I*PolyLog[2,-I*Sqrt[1+I*a*x]/Sqrt[1-I*a*x]]/(a*Sqrt[c]) -
  I*PolyLog[2,I*Sqrt[1+I*a*x]/Sqrt[1-I*a*x]]/(a*Sqrt[c]) /;
FreeQ[{a,c,d},x] && ZeroQ[d-a^2*c] && PositiveQ[c]
```

- **Basis:** $\partial_x \frac{\sqrt{1+a^2 x^2}}{\sqrt{c+c\,a^2 x^2}} = 0$

- **Rule: If** $d = a^2 c \wedge \neg (c > 0)$, **then**

$$\int \frac{\mathrm{ArcTan}[a\,x]}{\sqrt{c + d\,x^2}}\,dx \longrightarrow \frac{\sqrt{1 + a^2 x^2}}{\sqrt{c + d\,x^2}} \int \frac{\mathrm{ArcTan}[a\,x]}{\sqrt{1 + a^2 x^2}}\,dx$$

- **Program code:**

```
Int[ArcTan[a_.*x_]/Sqrt[c_+d_.*x_^2],x_Symbol] :=
  Sqrt[1+a^2*x^2]/Sqrt[c+d*x^2]*Int[ArcTan[a*x]/Sqrt[1+a^2*x^2],x] /;
FreeQ[{a,c,d},x] && ZeroQ[d-a^2*c] && Not[PositiveQ[c]]
```

- **Rule: If** $d = a^2 c$, **then**

$$\int \frac{\mathrm{ArcTan}[a\,x]}{\left(c + d\,x^2\right)^{3/2}}\,dx \longrightarrow \frac{1}{a\,c\sqrt{c + d\,x^2}} + \frac{x\,\mathrm{ArcTan}[a\,x]}{c\sqrt{c + d\,x^2}}$$

- **Program code:**

```
Int[ArcTan[a_.*x_]/(c_+d_.*x_^2)^(3/2),x_Symbol] :=
  1/(a*c*Sqrt[c+d*x^2]) +
  x*ArcTan[a*x]/(c*Sqrt[c+d*x^2]) /;
FreeQ[{a,c,d},x] && ZeroQ[d-a^2*c]
```

- **Rule: If** $d = a^2 c \wedge n > 1$**, then**

$$\int \frac{\text{ArcTan}[a\,x]^n}{\left(c + d\,x^2\right)^{3/2}}\,dx \longrightarrow \frac{n\,\text{ArcTan}[a\,x]^{n-1}}{a\,c\,\sqrt{c + d\,x^2}} + \frac{x\,\text{ArcTan}[a\,x]^n}{c\,\sqrt{c + d\,x^2}} - n\,(n-1)\int \frac{\text{ArcTan}[a\,x]^{n-2}}{\left(c + d\,x^2\right)^{3/2}}\,dx$$

- **Program code:**

```
Int[ArcTan[a_.*x_]^n_/(c_+d_.*x_^2)^(3/2),x_Symbol] :=
  n*ArcTan[a*x]^(n-1)/(a*c*Sqrt[c+d*x^2]) +
  x*ArcTan[a*x]^n/(c*Sqrt[c+d*x^2]) -
  Dist[n*(n-1),Int[ArcTan[a*x]^(n-2)/(c+d*x^2)^(3/2),x]] /;
FreeQ[{a,c,d},x] && ZeroQ[d-a^2*c] && RationalQ[n] && n>1
```

- **Rule: If** $d = a^2 c \wedge n < -1 \wedge n \neq -2$**, then**

$$\int \frac{\text{ArcTan}[a\,x]^n}{\left(c + d\,x^2\right)^{3/2}}\,dx \longrightarrow$$

$$\frac{\text{ArcTan}[a\,x]^{n+1}}{a\,c\,(n+1)\,\sqrt{c + d\,x^2}} + \frac{x\,\text{ArcTan}[a\,x]^{n+2}}{c\,(n+1)\,(n+2)\,\sqrt{c + d\,x^2}} - \frac{1}{(n+1)\,(n+2)}\int \frac{\text{ArcTan}[a\,x]^{n+2}}{\left(c + d\,x^2\right)^{3/2}}\,dx$$

- **Program code:**

```
Int[ArcTan[a_.*x_]^n_/(c_+d_.*x_^2)^(3/2),x_Symbol] :=
  ArcTan[a*x]^(n+1)/(a*c*(n+1)*Sqrt[c+d*x^2]) +
  x*ArcTan[a*x]^(n+2)/(c*(n+1)*(n+2)*Sqrt[c+d*x^2]) -
  Dist[1/((n+1)*(n+2)),Int[ArcTan[a*x]^(n+2)/(c+d*x^2)^(3/2),x]] /;
FreeQ[{a,c,d},x] && ZeroQ[d-a^2*c] && RationalQ[n] && n<-1 && n≠-2
```

- **Rule: If** $d = a^2 c \wedge m > 0$**, then**

$$\int \left(c + d\,x^2\right)^m \text{ArcTan}[a\,x]\,dx \longrightarrow$$

$$-\frac{\left(c + d\,x^2\right)^m}{2\,a\,m\,(2\,m+1)} + \frac{x\,\left(c + d\,x^2\right)^m \text{ArcTan}[a\,x]}{(2\,m+1)} + \frac{2\,c\,m}{2\,m+1}\int \left(c + d\,x^2\right)^{m-1} \text{ArcTan}[a\,x]\,dx$$

- **Program code:**

```
Int[(c_+d_.*x_^2)^m_.*ArcTan[a_.*x_],x_Symbol] :=
  -(c+d*x^2)^m/(2*a*m*(2*m+1)) +
  x*(c+d*x^2)^m*ArcTan[a*x]/(2*m+1) +
  Dist[2*c*m/(2*m+1),Int[(c+d*x^2)^(m-1)*ArcTan[a*x],x]] /;
FreeQ[{a,c,d},x] && ZeroQ[d-a^2*c] && RationalQ[m] && m>0
```

- **Rule: If** $d = a^2 c \bigwedge m < -1 \bigwedge m \neq -\frac{3}{2}$**, then**

$$\int \left(c + d\,x^2\right)^m \text{ArcTan}[a\,x]\,dx \longrightarrow$$

$$\frac{\left(c + d\,x^2\right)^{m+1}}{4\,a\,c\,(m+1)^2} - \frac{x\left(c + d\,x^2\right)^{m+1} \text{ArcTan}[a\,x]}{2\,c\,(m+1)} + \frac{2\,m+3}{2\,c\,(m+1)} \int \left(c + d\,x^2\right)^{m+1} \text{ArcTan}[a\,x]\,dx$$

- **Program code:**

```
Int[(c_+d_.*x_^2)^m_*ArcTan[a_.*x_],x_Symbol] :=
  (c+d*x^2)^(m+1)/(4*a*c*(m+1)^2) -
  x*(c+d*x^2)^(m+1)*ArcTan[a*x]/(2*c*(m+1)) +
  Dist[(2*m+3)/(2*c*(m+1)),Int[(c+d*x^2)^(m+1)*ArcTan[a*x],x]] /;
FreeQ[{a,c,d},x] && ZeroQ[d-a^2*c] && RationalQ[m] && m<-1 && m≠-3/2
```

- **Rule: If** $d = a^2 c \bigwedge m < -1 \bigwedge m \neq -\frac{3}{2} \bigwedge n > 1$**, then**

$$\int \left(c + d\,x^2\right)^m \text{ArcTan}[a\,x]^n\,dx \longrightarrow$$

$$\frac{n\left(c + d\,x^2\right)^{m+1} \text{ArcTan}[a\,x]^{n-1}}{4\,a\,c\,(m+1)^2} - \frac{x\left(c + d\,x^2\right)^{m+1} \text{ArcTan}[a\,x]^n}{2\,c\,(m+1)} +$$
$$\frac{2\,m+3}{2\,c\,(m+1)} \int \left(c + d\,x^2\right)^{m+1} \text{ArcTan}[a\,x]^n\,dx - \frac{n\,(n-1)}{4\,(m+1)^2} \int \left(c + d\,x^2\right)^m \text{ArcTan}[a\,x]^{n-2}\,dx$$

- **Program code:**

```
Int[(c_+d_.*x_^2)^m_*ArcTan[a_.*x_]^n_,x_Symbol] :=
  n*(c+d*x^2)^(m+1)*ArcTan[a*x]^(n-1)/(4*a*c*(m+1)^2) -
  x*(c+d*x^2)^(m+1)*ArcTan[a*x]^n/(2*c*(m+1)) +
  Dist[(2*m+3)/(2*c*(m+1)),Int[(c+d*x^2)^(m+1)*ArcTan[a*x]^n,x]] -
  Dist[n*(n-1)/(4*(m+1)^2),Int[(c+d*x^2)^m*ArcTan[a*x]^(n-2),x]] /;
FreeQ[{a,c,d},x] && ZeroQ[d-a^2*c] && RationalQ[{m,n}] && m<-1 && m≠-3/2 && n>1
```

- **Derivation: Integration by parts**
- **Rule: If** $d = a^2 c \land m < -1 \land n < -1$**, then**

$$\int (c + d x^2)^m \operatorname{ArcTan}[a x]^n \, dx \longrightarrow$$

$$\frac{(c + d x^2)^{m+1} \operatorname{ArcTan}[a x]^{n+1}}{a c (n+1)} - \frac{2 a (m+1)}{n+1} \int x (c + d x^2)^m \operatorname{ArcTan}[a x]^{n+1} \, dx$$

- **Program code:**

```
Int[(c_+d_.*x_^2)^m_*ArcTan[a_.*x_]^n_,x_Symbol] :=
  (c+d*x^2)^(m+1)*ArcTan[a*x]^(n+1)/(a*c*(n+1)) -
  Dist[2*a*(m+1)/(n+1),Int[x*(c+d*x^2)^m*ArcTan[a*x]^(n+1),x]] /;
FreeQ[{a,c,d},x] && ZeroQ[d-a^2*c] && RationalQ[{m,n}] && m<-1 && n<-1
```

- **Derivation: Integration by substitution**
- **Basis:** $(1 + a^2 x^2)^m \operatorname{ArcTan}[a x]^n = \frac{1}{a} \operatorname{Sec}[\operatorname{ArcTan}[a x]]^{2 (m+1)} \operatorname{ArcTan}[a x]^n \partial_x \operatorname{ArcTan}[a x]$
- **Rule: If** $d = a^2 c \land m, n \in \mathbb{Q} \land m < -1 \land (n < 0 \lor n \notin \mathbb{Z}) \land (m \in \mathbb{Z} \lor c > 0)$**, then**

$$\int (c + d x^2)^m \operatorname{ArcTan}[a x]^n \, dx \longrightarrow \frac{c^m}{a} \operatorname{Subst}\left[\int x^n \operatorname{Sec}[x]^{2 (m+1)} \, dx, x, \operatorname{ArcTan}[a x]\right]$$

- **Program code:**

```
Int[(c_+d_.*x_^2)^m_*ArcTan[a_.*x_]^n_,x_Symbol] :=
  Dist[c^m/a,Subst[Int[x^n*Sec[x]^(2*(m+1)),x],x,ArcTan[a*x]]] /;
FreeQ[{a,c,d},x] && ZeroQ[d-a^2*c] && RationalQ[{m,n}] && m<-1 && (n<0 || Not[IntegerQ[n]]) && (Inte
```

- **Basis: If** $d = a^2 c$, $D\left[\frac{c^{m-\frac{1}{2}} \sqrt{c+d x^2}}{\sqrt{1+a^2 x^2}}, x\right] = 0$
- **Rule: If** $d = a^2 c \land m, n \in \mathbb{Q} \land m < -1 \land (n < 0 \lor n \notin \mathbb{Z}) \land m - \frac{1}{2} \in \mathbb{Z} \land \neg (c > 0)$**, then**

$$\int (c + d x^2)^m \operatorname{ArcTan}[a x]^n \, dx \longrightarrow \frac{c^{m-\frac{1}{2}} \sqrt{c + d x^2}}{\sqrt{1 + a^2 x^2}} \int (1 + a^2 x^2)^m \operatorname{ArcTan}[a x]^n \, dx$$

- **Program code:**

```
Int[(c_+d_.*x_^2)^m_*ArcTan[a_.*x_]^n_,x_Symbol] :=
  c^(m-1/2)*Sqrt[c+d*x^2]/Sqrt[1+a^2*x^2]*Int[(1+a^2*x^2)^m*ArcTan[a*x]^n,x] /;
FreeQ[{a,c,d},x] && ZeroQ[d-a^2*c] && RationalQ[{m,n}] && m<-1 && (n<0 || Not[IntegerQ[n]]) && Integ
```

$$\int x^m \left(c + d\,x^2\right)^p \operatorname{ArcTan}[a\,x]^n \,dx$$

- **Derivation: Integration by parts**

- **Rule: If** $d = a^2 c \wedge p \in \mathbb{Q} \wedge n > 0 \wedge p \neq -1$**, then**

$$\int x \left(c + d\,x^2\right)^p \operatorname{ArcTan}[a\,x]^n \,dx \longrightarrow \frac{\left(c + d\,x^2\right)^{p+1} \operatorname{ArcTan}[a\,x]^n}{2\,d\,(p+1)} - \frac{n}{2\,a\,(p+1)} \int \left(c + d\,x^2\right)^p \operatorname{ArcTan}[a\,x]^{n-1} \,dx$$

- **Program code:**

```
Int[x_*(c_+d_.*x_^2)^p_.*ArcTan[a_.*x_]^n_.,x_Symbol] :=
  (c+d*x^2)^(p+1)*ArcTan[a*x]^n/(2*d*(p+1)) -
  Dist[n/(2*a*(p+1)),Int[(c+d*x^2)^p*ArcTan[a*x]^(n-1),x]] /;
FreeQ[{a,c,d},x] && ZeroQ[d-a^2*c] && RationalQ[{n,p}] && n>0 && p≠-1
```

- **Rule: If** $d = a^2 c \wedge p \in \mathbb{Q}$**, then**

$$\int \frac{x \left(c + d\,x^2\right)^p}{\operatorname{ArcTan}[a\,x]^2} \,dx \longrightarrow -\frac{x \left(c + d\,x^2\right)^{p+1}}{a\,c \operatorname{ArcTan}[a\,x]} + \frac{1}{a} \int \frac{\left(1 + (2\,p+3)\,a^2 x^2\right) \left(c + d\,x^2\right)^p}{\operatorname{ArcTan}[a\,x]} \,dx$$

- **Program code:**

```
Int[x_*(c_+d_.*x_^2)^p_./ArcTan[a_.*x_]^2,x_Symbol] :=
  -x*(c+d*x^2)^(p+1)/(a*c*ArcTan[a*x]) +
  Dist[1/a,Int[(1+(2*p+3)*a^2*x^2)*(c+d*x^2)^p/ArcTan[a*x],x]] /;
FreeQ[{a,c,d},x] && ZeroQ[d-a^2*c] && RationalQ[p]
```

- **Rule: If** $d = a^2 c \wedge n < -1 \wedge n \neq -2$**, then**

$$\int \frac{x \operatorname{ArcTan}[a\,x]^n}{\left(c + d\,x^2\right)^2} \,dx \longrightarrow$$

$$\frac{x \operatorname{ArcTan}[a\,x]^{n+1}}{a\,c\,(n+1)\left(c + d\,x^2\right)} - \frac{\left(1 - a^2 x^2\right) \operatorname{ArcTan}[a\,x]^{n+2}}{d\,(n+1)\,(n+2)\left(c + d\,x^2\right)} - \frac{4}{(n+1)\,(n+2)} \int \frac{x \operatorname{ArcTan}[a\,x]^{n+2}}{\left(c + d\,x^2\right)^2} \,dx$$

- **Program code:**

```
Int[x_*ArcTan[a_.*x_]^n_/(c_+d_.*x_^2)^2,x_Symbol] :=
  x*ArcTan[a*x]^(n+1)/(a*c*(n+1)*(c+d*x^2)) -
  (1-a^2*x^2)*ArcTan[a*x]^(n+2)/(d*(n+1)*(n+2)*(c+d*x^2)) -
  Dist[4/((n+1)*(n+2)),Int[x*ArcTan[a*x]^(n+2)/(c+d*x^2)^2,x]] /;
FreeQ[{a,c,d},x] && ZeroQ[d-a^2*c] && RationalQ[n] && n<-1 && n≠-2
```

- **Derivation: Integration by parts**

- **Rule: If** $d = a^2 c \land m, n, 2p \in \mathbb{Z} \land m < -1 \land n > 0 \land m + 2p + 3 = 0$, **then**

$$\int x^m \left(c + d x^2\right)^p \text{ArcTan}[a x]^n \, dx \longrightarrow$$

$$\frac{x^{m+1} \left(c + d x^2\right)^{p+1} \text{ArcTan}[a x]^n}{c (m+1)} - \frac{a n}{m+1} \int x^{m+1} \left(c + d x^2\right)^p \text{ArcTan}[a x]^{n-1} \, dx$$

- **Program code:**

```
Int[x_^m_*(c_+d_.*x_^2)^p_.*ArcTan[a_.*x_]^n_.,x_Symbol] :=
  x^(m+1)*(c+d*x^2)^(p+1)*ArcTan[a*x]^n/(c*(m+1)) -
  Dist[a*n/(m+1),Int[x^(m+1)*(c+d*x^2)^p*ArcTan[a*x]^(n-1),x]] /;
FreeQ[{a,c,d},x] && ZeroQ[d-a^2*c] && IntegersQ[m,n,2*p] && m<-1 && n>0 && ZeroQ[m+2*p+3]
```

- **Derivation: Integration by parts**

- **Rule: If** $d = a^2 c \land m, n, 2p \in \mathbb{Z} \land n < -1 \land m + 2p + 2 = 0$, **then**

$$\int x^m \left(c + d x^2\right)^p \text{ArcTan}[a x]^n \, dx \longrightarrow$$

$$\frac{x^m \left(c + d x^2\right)^{p+1} \text{ArcTan}[a x]^{n+1}}{a c (n+1)} - \frac{m}{a (n+1)} \int x^{m-1} \left(c + d x^2\right)^p \text{ArcTan}[a x]^{n+1} \, dx$$

- **Program code:**

```
Int[x_^m_*(c_+d_.*x_^2)^p_.*ArcTan[a_.*x_]^n_.,x_Symbol] :=
  x^m*(c+d*x^2)^(p+1)*ArcTan[a*x]^(n+1)/(a*c*(n+1)) -
  Dist[m/(a*(n+1)),Int[x^(m-1)*(c+d*x^2)^p*ArcTan[a*x]^(n+1),x]] /;
FreeQ[{a,c,d},x] && ZeroQ[d-a^2*c] && IntegersQ[m,n,2*p] && n<-1 && ZeroQ[m+2*p+2]
```

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{x^2}{c+d x^2} = \frac{1}{d} - \frac{c}{d\,(c+d x^2)}$
- **Rule: If** $d = a^2 c \land m, n, 2p \in \mathbb{Z} \land m > 1 \land n \neq -1 \land p < -1$**, then**

$$\int x^m \left(c + d x^2\right)^p \text{ArcTan}[a x]^n \, dx \longrightarrow \frac{1}{d} \int x^{m-2} \left(c + d x^2\right)^{p+1} \text{ArcTan}[a x]^n \, dx - \frac{c}{d} \int x^{m-2} \left(c + d x^2\right)^p \text{ArcTan}[a x]^n \, dx$$

- **Program code:**

```
Int[x_^m_*(c_+d_.*x_^2)^p_*ArcTan[a_.*x_]^n_.,x_Symbol] :=
  Dist[1/d,Int[x^(m-2)*(c+d*x^2)^(p+1)*ArcTan[a*x]^n,x]] -
  Dist[c/d,Int[x^(m-2)*(c+d*x^2)^p*ArcTan[a*x]^n,x]] /;
FreeQ[{a,c,d},x] && ZeroQ[d-a^2*c] && IntegersQ[m,n,2*p] && m>1 && n≠-1 && p<-1
```

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{1}{c+d x^2} = \frac{1}{c} - \frac{d x^2}{c\,(c+d x^2)}$
- **Rule: If** $d = a^2 c \land m, n, 2p \in \mathbb{Z} \land m < 0 \land n \neq -1 \land p < -1$**, then**

$$\int x^m \left(c + d x^2\right)^p \text{ArcTan}[a x]^n \, dx \longrightarrow \frac{1}{c} \int x^m \left(c + d x^2\right)^{p+1} \text{ArcTan}[a x]^n \, dx - \frac{d}{c} \int x^{m+2} \left(c + d x^2\right)^p \text{ArcTan}[a x]^n \, dx$$

- **Program code:**

```
Int[x_^m_*(c_+d_.*x_^2)^p_*ArcTan[a_.*x_]^n_.,x_Symbol] :=
  Dist[1/c,Int[x^m*(c+d*x^2)^(p+1)*ArcTan[a*x]^n,x]] -
  Dist[d/c,Int[x^(m+2)*(c+d*x^2)^p*ArcTan[a*x]^n,x]] /;
FreeQ[{a,c,d},x] && ZeroQ[d-a^2*c] && IntegersQ[m,n,2*p] && m<0 && n≠-1 && p<-1
```

- **Derivation: Integration by parts**

- **Rule: If** $d = a^2 c \land m, n, 2p \in \mathbb{Z} \land m < -1 \land n > 0 \land m + 2p + 3 \neq 0$, **then**

$$\int x^m \left(c + d x^2\right)^p \text{ArcTan}[a x]^n \, dx \longrightarrow \frac{x^{m+1} \left(c + d x^2\right)^{p+1} \text{ArcTan}[a x]^n}{c (m+1)} - \frac{a n}{m+1} \int x^{m+1} \left(c + d x^2\right)^p \text{ArcTan}[a x]^{n-1} \, dx - \frac{a^2 (m + 2p + 3)}{m+1} \int x^{m+2} \left(c + d x^2\right)^p \text{ArcTan}[a x]^n \, dx$$

- **Program code:**

```
Int[x_^m_*(c_+d_.*x_^2)^p_.*ArcTan[a_.*x_]^n_.,x_Symbol] :=
  x^(m+1)*(c+d*x^2)^(p+1)*ArcTan[a*x]^n/(c*(m+1)) -
  Dist[a*n/(m+1),Int[x^(m+1)*(c+d*x^2)^p*ArcTan[a*x]^(n-1),x]] -
  Dist[a^2*(m+2*p+3)/(m+1),Int[x^(m+2)*(c+d*x^2)^p*ArcTan[a*x]^n,x]] /;
FreeQ[{a,c,d},x] && ZeroQ[d-a^2*c] && IntegersQ[m,n,2*p] && m<-1 && n>0 && NonzeroQ[m+2*p+3]
```

- **Derivation: Integration by parts**

- **Rule: If** $d = a^2 c \land m, n, 2p \in \mathbb{Z} \land n < -1 \land m + 2p + 2 \neq 0$, **then**

$$\int x^m \left(c + d x^2\right)^p \text{ArcTan}[a x]^n \, dx \longrightarrow \frac{x^m \left(c + d x^2\right)^{p+1} \text{ArcTan}[a x]^{n+1}}{a c (n+1)} - \frac{m}{a (n+1)} \int x^{m-1} \left(c + d x^2\right)^p \text{ArcTan}[a x]^{n+1} \, dx - \frac{a (m + 2p + 2)}{n+1} \int x^{m+1} \left(c + d x^2\right)^p \text{ArcTan}[a x]^{n+1} \, dx$$

- **Program code:**

```
Int[x_^m_.*(c_+d_.*x_^2)^p_.*ArcTan[a_.*x_]^n_.,x_Symbol] :=
  x^m*(c+d*x^2)^(p+1)*ArcTan[a*x]^(n+1)/(a*c*(n+1)) -
  Dist[m/(a*(n+1)),Int[x^(m-1)*(c+d*x^2)^p*ArcTan[a*x]^(n+1),x]] -
  Dist[a*(m+2*p+2)/(n+1),Int[x^(m+1)*(c+d*x^2)^p*ArcTan[a*x]^(n+1),x]] /;
FreeQ[{a,c,d},x] && ZeroQ[d-a^2*c] && IntegersQ[m,n,2*p] && n<-1 && NonzeroQ[m+2*p+2]
```

- **Derivation: Integration by substitution**

- **Basis: If** $m \in \mathbb{Z}$ **or** $a > 0$, $(e + f x^m) (1 + a^2 x^2)^p \text{ArcTan}[a x]^n =$
$\frac{1}{a^{m+1}} (e a^m + f \text{Tan}[\text{ArcTan}[a x]]^m) \text{Sec}[\text{ArcTan}[a x]]^{2 (p+1)} \text{ArcTan}[a x]^n \partial_x \text{ArcTan}[a x]$

- **Rule: If** $d = a^2 c \land m, n, p \in \mathbb{Q} \land p < -1 \land (n < 0 \lor n \notin \mathbb{Z}) \land (p \in \mathbb{Z} \lor c > 0) \land (m \in \mathbb{Z} \lor a > 0)$, **then**

$$\int (e + f x^m) \left(c + d x^2\right)^p \text{ArcTan}[a x]^n \, dx \longrightarrow \frac{c^p}{a^{m+1}}$$
$$\text{Subst}\left[\int x^n (e a^m + f \text{Tan}[x]^m) \text{Sec}[x]^{2 (p+1)} \, dx, x, \text{ArcTan}[a x]\right]$$

- **Program code:**

```
Int[(e_.+f_.*x_^m_.)*(c_+d_.*x_^2)^p_*ArcTan[a_.*x_]^n_,x_Symbol] :=
  Dist[c^p/a^(m+1),Subst[Int[Expand[x^n*TrigReduce[Regularize[(e*a^m+f*Tan[x]^m)*Sec[x]^(2*(p+1)),x]
FreeQ[{a,c,d,e,f},x] && ZeroQ[d-a^2*c] && RationalQ[{m,n,p}] && p<-1 && (n<0 || Not[IntegerQ[n]]) &&
```

- **Derivation: Integration by substitution**

- **Basis:** $x^m (1 + a^2 x^2)^p \text{ArcTan}[a x]^n = \frac{1}{a} \left(\frac{\text{Tan}[\text{ArcTan}[a x]]}{a}\right)^m \text{Sec}[\text{ArcTan}[a x]]^{2 (p+1)} \text{ArcTan}[a x]^n \partial_x \text{ArcTan}[a x]$

- **Rule: If** $d = a^2 c \land m, n \in \mathbb{Q} \land p < -1 \land (n < 0 \lor n \notin \mathbb{Z}) \land (p \in \mathbb{Z} \lor c > 0) \land \neg (m \in \mathbb{Z} \lor a > 0)$, **then**

$$\int x^m \left(c + d x^2\right)^p \text{ArcTan}[a x]^n \, dx \longrightarrow \frac{c^p}{a} \text{Subst}\left[\int x^n (\text{Tan}[x] / a)^m \text{Sec}[x]^{2 (p+1)} \, dx, x, \text{ArcTan}[a x]\right]$$

- **Program code:**

```
Int[x_^m_.*(c_+d_.*x_^2)^p_*ArcTan[a_.*x_]^n_,x_Symbol] :=
  Dist[c^p/a,Subst[Int[x^n*(Tan[x]/a)^m*Sec[x]^(2*(p+1)),x],x,ArcTan[a*x]]] /;
FreeQ[{a,c,d},x] && ZeroQ[d-a^2*c] && RationalQ[{m,n,p}] && p<-1 && (n<0 || Not[IntegerQ[n]]) && (In
```

- **Basis: If** $d = a^2 c$, $D\left[\frac{c^{p-\frac{1}{2}} \sqrt{c+d x^2}}{\sqrt{1+a^2 x^2}}, x\right] = 0$

- **Rule: If** $d = a^2 c \land m, n \in \mathbb{Q} \bigwedge p < -1 \bigwedge (n < 0 \lor n \notin \mathbb{Z}) \bigwedge p - \frac{1}{2} \in \mathbb{Z} \bigwedge \neg (c > 0)$, **then**

$$\int x^m \left(c + d x^2\right)^p \text{ArcTan}[a x]^n \, dx \longrightarrow \frac{c^{p-\frac{1}{2}} \sqrt{c + d x^2}}{\sqrt{1 + a^2 x^2}} \int x^m \left(1 + a^2 x^2\right)^p \text{ArcTan}[a x]^n \, dx$$

- **Program code:**

```
Int[x_^m_.*(c_+d_.*x_^2)^p_*ArcTan[a_.*x_]^n_,x_Symbol] :=
  c^(p-1/2)*Sqrt[c+d*x^2]/Sqrt[1+a^2*x^2]*Int[x^m*(1+a^2*x^2)^p*ArcTan[a*x]^n,x] /;
FreeQ[{a,c,d},x] && ZeroQ[d-a^2*c] && RationalQ[{m,n,p}] && p<-1 && (n<0 || Not[IntegerQ[n]]) && Int
```

$$\int \text{ArcTan}[a + b\,x^n]\,dx$$

- **Reference: G&R 2.822.1, CRC 443, A&S 4.4.60**
- **Derivation: Integration by parts**
- **Rule:**

$$\int \text{ArcTan}[a + b\,x]\,dx \longrightarrow \frac{(a + b\,x)\,\text{ArcTan}[a + b\,x]}{b} - \frac{\text{Log}\left[1 + (a + b\,x)^2\right]}{2\,b}$$

- **Program code:**

```
Int[ArcTan[a_+b_.*x_],x_Symbol] :=
  (a+b*x)*ArcTan[a+b*x]/b - Log[1+(a+b*x)^2]/(2*b) /;
FreeQ[{a,b},x]
```

- **Reference: G&R 2.822.1, CRC 443, A&S 4.4.60**
- **Derivation: Integration by parts**
- **Rule: If $n \in \mathbb{Q}$, then**

$$\int \text{ArcTan}[a + b\,x^n]\,dx \longrightarrow x\,\text{ArcTan}[a + b\,x^n] - b\,n \int \frac{x^n}{1 + a^2 + 2\,a\,b\,x^n + b^2\,x^{2\,n}}\,dx$$

- **Program code:**

```
Int[ArcTan[a_+b_.*x_^n_],x_Symbol] :=
  x*ArcTan[a+b*x^n] -
  Dist[b*n,Int[x^n/(1+a^2+2*a*b*x^n+b^2*x^(2*n)),x]] /;
FreeQ[{a,b},x] && RationalQ[n]
```

$$\int x^m \operatorname{ArcTan}[a+b\,x^n]\,dx$$

- **Derivation: Algebraic expansion**
- **Basis:** $\operatorname{ArcTan}[z] = \frac{1}{2}\,i\operatorname{Log}[1-i\,z] - \frac{1}{2}\,i\operatorname{Log}[1+i\,z]$
- **Rule:**

$$\int \frac{\operatorname{ArcTan}[a+b\,x^n]}{x}\,dx \longrightarrow \frac{I}{2}\int \frac{\operatorname{Log}[1-I\,a-I\,b\,x^n]}{x}\,dx - \frac{I}{2}\int \frac{\operatorname{Log}[1+I\,a+I\,b\,x^n]}{x}\,dx$$

- **Program code:**

```
Int[ArcTan[a_.+b_.*x_^n_.]/x_,x_Symbol] :=
  Dist[I/2,Int[Log[1-I*a-I*b*x^n]/x,x]] -
  Dist[I/2,Int[Log[1+I*a+I*b*x^n]/x,x]] /;
FreeQ[{a,b,n},x]
```

- **Reference: G&R 2.851, CRC 456, A&S 4.4.69**
- **Derivation: Integration by parts**
- **Rule: If** $m, n \in \mathbb{Q} \land m+1 \neq 0 \land m+1 \neq n$**, then**

$$\int x^m \operatorname{ArcTan}[a+b\,x^n]\,dx \longrightarrow \frac{x^{m+1}\operatorname{ArcTan}[a+b\,x^n]}{m+1} - \frac{b\,n}{m+1}\int \frac{x^{m+n}}{1+a^2+2\,a\,b\,x^n+b^2\,x^{2\,n}}\,dx$$

- **Program code:**

```
Int[x_^m_.*ArcTan[a_+b_.*x_^n_.],x_Symbol] :=
  x^(m+1)*ArcTan[a+b*x^n]/(m+1) -
  Dist[b*n/(m+1),Int[x^(m+n)/(1+a^2+2*a*b*x^n+b^2*x^(2*n)),x]] /;
FreeQ[{a,b},x] && RationalQ[{m,n}] && m+1≠0 && m+1≠n
```

$$\int \text{ArcTan}[a + b\,x]^n\,dx$$

- **Derivation: Integration by substitution**
- **Rule: If** $n \in \mathbb{Z} \land n > 1$**, then**

$$\int \text{ArcTan}[a + b\,x]^n\,dx \longrightarrow \frac{1}{b}\,\text{Subst}\Big[\int \text{ArcTan}[x]^n\,dx,\,x,\,a + b\,x\Big]$$

- **Program code:**

```
Int[ArcTan[a_+b_.*x_]^n_,x_Symbol] :=
  Dist[1/b,Subst[Int[ArcTan[x]^n,x],x,a+b*x]] /;
FreeQ[{a,b},x] && IntegerQ[n] && n>1
```

$$\int x^m \operatorname{ArcTan}[a + b x]^n \, dx$$

- **Derivation: Integration by substitution**

- **Rule: If $m, n \in \mathbb{Z} \land m > 0 \land n > 1$, then**

$$\int x^m \operatorname{ArcTan}[a + b x]^n \, dx \longrightarrow \frac{1}{b^{m+1}} \operatorname{Subst}\left[\int (x - a)^m \operatorname{ArcTan}[x]^n \, dx,\ x,\ a + b x\right]$$

- **Program code:**

```
Int[x_^m_.*ArcTan[a_+b_.*x_]^n_,x_Symbol] :=
  Dist[1/b^(m+1),Subst[Int[(x-a)^m*ArcTan[x]^n,x],x,a+b*x]] /;
FreeQ[{a,b},x] && IntegersQ[m,n] && m>0 && n>1
```

$$\int \frac{\text{ArcTan}[a + b\,x]}{c + d\,x^n}\,dx$$

- **Derivation: Algebraic simplification**
- **Basis:** $\text{ArcTan}[z] = \frac{1}{2}\,i\,\text{Log}[1 - i\,z] - \frac{1}{2}\,i\,\text{Log}[1 + i\,z]$
- **Rule: If** $n \in \mathbb{Z} \land \neg\,(n = 2 \land d = b^2\,c)$, **then**

$$\int \frac{\text{ArcTan}[b\,x]}{c + d\,x^n}\,dx \longrightarrow \frac{I}{2}\int \frac{\text{Log}[1 - I\,b\,x]}{c + d\,x^n}\,dx - \frac{I}{2}\int \frac{\text{Log}[1 + I\,b\,x]}{c + d\,x^n}\,dx$$

- **Program code:**

```
Int[ArcTan[b_.*x_]/(c_+d_.*x_^n_.),x_Symbol] :=
  Dist[I/2,Int[Log[1-I*b*x]/(c+d*x^n),x]] -
  Dist[I/2,Int[Log[1+I*b*x]/(c+d*x^n),x]] /;
FreeQ[{b,c,d},x] && IntegerQ[n] && Not[n==2 && ZeroQ[d-b^2*c]]
```

- **Derivation: Algebraic simplification**
- **Basis:** $\text{ArcTan}[z] = \frac{1}{2}\,i\,\text{Log}[1 - i\,z] - \frac{1}{2}\,i\,\text{Log}[1 + i\,z]$
- **Rule: If** $n \in \mathbb{Z} \land \neg\,(n = 1 \land a\,d - b\,c = 0)$, **then**

$$\int \frac{\text{ArcTan}[a + b\,x]}{c + d\,x^n}\,dx \longrightarrow \frac{I}{2}\int \frac{\text{Log}[1 - I\,a - I\,b\,x]}{c + d\,x^n}\,dx - \frac{I}{2}\int \frac{\text{Log}[1 + I\,a + I\,b\,x]}{c + d\,x^n}\,dx$$

- **Program code:**

```
Int[ArcTan[a_+b_.*x_]/(c_+d_.*x_^n_.),x_Symbol] :=
  Dist[I/2,Int[Log[1-I*a-I*b*x]/(c+d*x^n),x]] -
  Dist[I/2,Int[Log[1+I*a+I*b*x]/(c+d*x^n),x]] /;
FreeQ[{a,b,c,d},x] && IntegerQ[n] && Not[n==1 && ZeroQ[a*d-b*c]]
```

$$\int u \operatorname{ArcTan}\left[\frac{c}{a + b x^n}\right]^m dx$$

- **Derivation: Algebraic simplification**
- **Basis:** $\operatorname{ArcTan}[z] = \operatorname{ArcCot}\left[\frac{1}{z}\right]$
- **Rule:**

$$\int u \operatorname{ArcTan}\left[\frac{c}{a + b x^n}\right]^m dx \longrightarrow \int u \operatorname{ArcCot}\left[\frac{a}{c} + \frac{b x^n}{c}\right]^m dx$$

- **Program code:**

```
Int[u_.*ArcTan[c_./(a_.+b_.*x_^n_.)]^m_.,x_Symbol] :=
  Int[u*ArcCot[a/c+b*x^n/c]^m,x] /;
FreeQ[{a,b,c,n,m},x]
```

$$\int \frac{f[x,\ \text{ArcTan}[a+b\,x]]}{1-(a+b\,x)^2}\,dx$$

- **Derivation: Integration by substitution**

- **Basis:** $\frac{f[z]}{1+z^2} = f[\text{Tan}[\text{ArcTan}[z]]]\ \text{ArcTan}'[z]$

- **Basis:** $r+s\,x+t\,x^2 = -\frac{s^2-4\,r\,t}{4\,t}\left(1-\frac{(s+2\,t\,x)^2}{s^2-4\,r\,t}\right)$

- **Basis:** $1+\text{Tan}[z]^2 = \text{Sec}[z]^2$

- **Rule:**

$$\int \frac{f[x,\ \text{ArcTan}[a+b\,x]]}{1+(a+b\,x)^2}\,dx \longrightarrow \frac{1}{b}\,\text{Subst}\left[\int f\left[-\frac{a}{b}+\frac{\text{Tan}[x]}{b},\ x\right]dx,\ x,\ \text{ArcTan}[a+b\,x]\right]$$

- **Program code:**

```
If[ShowSteps,

Int[u_*v_^n_.,x_Symbol] :=
  Module[{tmp=InverseFunctionOfLinear[u,x]},
  ShowStep["","Int[f[x,ArcTan[a+b*x]]/(1+(a+b*x)^2),x]",
           "Subst[Int[f[-a/b+Tan[x]/b,x],x],x,ArcTan[a+b*x]]/b",Hold[
  Dist[(-Discriminant[v,x]/(4*Coefficient[v,x,2]))^n/Coefficient[tmp[[1]],x,1],
    Subst[Int[Regularize[SubstForInverseFunction[u,tmp,x]*Sec[x]^(2*(n+1)),x],x], x, tmp]]]] /;
 NotFalseQ[tmp] && Head[tmp]===ArcTan && ZeroQ[Discriminant[v,x]*tmp[[1]]^2+D[v,x]^2]] /;
SimplifyFlag && QuadraticQ[v,x] && IntegerQ[n] && n<0 && NegQ[Discriminant[v,x]] && MatchQ[u,r_.*f_^

Int[u_*v_^n_.,x_Symbol] :=
  Module[{tmp=InverseFunctionOfLinear[u,x]},
  Dist[(-Discriminant[v,x]/(4*Coefficient[v,x,2]))^n/Coefficient[tmp[[1]],x,1],
    Subst[Int[Regularize[SubstForInverseFunction[u,tmp,x]*Sec[x]^(2*(n+1)),x],x], x, tmp]] /;
 NotFalseQ[tmp] && Head[tmp]===ArcTan && ZeroQ[Discriminant[v,x]*tmp[[1]]^2+D[v,x]^2]] /;
QuadraticQ[v,x] && IntegerQ[n] && n<0 && NegQ[Discriminant[v,x]] && MatchQ[u,r_.*f_^w_ /; FreeQ[f,x]
```

$$\int \text{ArcTan}\left[a+b\,f^{c+d\,x}\right]\,dx$$

- **Derivation: Algebraic simplification**
- **Basis:** $\text{ArcTan}[z] = \frac{1}{2}\,i\,\text{Log}[1-i\,z] - \frac{1}{2}\,i\,\text{Log}[1+i\,z]$
- **Rule:**

$$\int \text{ArcTan}\left[a+b\,f^{c+d\,x}\right]\,dx \longrightarrow \frac{I}{2}\int \text{Log}\left[1-I\left(a+b\,f^{c+d\,x}\right)\right]\,dx - \frac{I}{2}\int \text{Log}\left[1+I\left(a+b\,f^{c+d\,x}\right)\right]\,dx$$

- **Program code:**

```
Int[ArcTan[a_.+b_.*f_^(c_.+d_.*x_)],x_Symbol] :=
  Dist[I/2,Int[Log[1-I*a-I*b*f^(c+d*x)],x]] -
  Dist[I/2,Int[Log[1+I*a+I*b*f^(c+d*x)],x]] /;
FreeQ[{a,b,c,d,f},x]
```

$$\int x^m \operatorname{ArcTan}\left[a + b\, f^{c+d\,x}\right] dx$$

- **Derivation: Algebraic simplification**

- **Basis:** $\operatorname{ArcTan}[z] = \frac{1}{2}\, i \operatorname{Log}[1 - i\, z] - \frac{1}{2}\, i \operatorname{Log}[1 + i\, z]$

- **Rule: If** $m \in \mathbb{Z} \wedge m > 0$**, then**

$$\int x^m \operatorname{ArcTan}\left[b\, f^{c+d\,x}\right] dx \longrightarrow \frac{I}{2} \int x^m \operatorname{Log}\left[1 - I\left(a + b\, f^{c+d\,x}\right)\right] dx - \frac{I}{2} \int x^m \operatorname{Log}\left[1 + I\left(a + b\, f^{c+d\,x}\right)\right] dx$$

- **Program code:**

```
Int[x_^m_.*ArcTan[a_.+b_.*f_^(c_.+d_.*x_)],x_Symbol] :=
  Dist[I/2,Int[x^m*Log[1-I*a-I*b*f^(c+d*x)],x]] -
  Dist[I/2,Int[x^m*Log[1+I*a+I*b*f^(c+d*x)],x]] /;
FreeQ[{a,b,c,d,f},x] && IntegerQ[m] && m>0
```

$$\int v\,\mathrm{ArcTan}[u]\,dx$$

- **Derivation: Integration by parts**
- **Rule: If u is free of inverse functions, then**

$$\int \mathrm{ArcTan}[u]\,dx \longrightarrow x\,\mathrm{ArcTan}[u] - \int \frac{x\,\partial_x u}{1+u^2}\,dx$$

- **Program code:**

```
Int[ArcTan[u_],x_Symbol] :=
  x*ArcTan[u] -
  Int[Regularize[x*D[u,x]/(1+u^2),x],x] /;
InverseFunctionFreeQ[u,x]
```

- **Derivation: Integration by parts**
- **Rule: If $m+1 \neq 0 \land$ u is free of inverse functions, then**

$$\int x^m\,\mathrm{ArcTan}[u]\,dx \longrightarrow \frac{x^{m+1}\,\mathrm{ArcTan}[u]}{m+1} - \frac{1}{m+1}\int \frac{x^{m+1}\,\partial_x u}{1+u^2}\,dx$$

- **Program code:**

```
Int[x_^m_.*ArcTan[u_],x_Symbol] :=
  x^(m+1)*ArcTan[u]/(m+1) -
  Dist[1/(m+1),Int[Regularize[x^(m+1)*D[u,x]/(1+u^2),x],x]] /;
FreeQ[m,x] && NonzeroQ[m+1] && InverseFunctionFreeQ[u,x] &&
    Not[FunctionOfQ[x^(m+1),u,x]] &&
    FalseQ[PowerVariableExpn[u,m+1,x]]
```

- **Derivation: Integration by parts**
- **Rule: If u is free of inverse functions, let $w = \int v \, dx$, if w is free of inverse functions, then**

$$\int v \operatorname{ArcTan}[u] \, dx \longrightarrow w \operatorname{ArcTan}[u] - \int \frac{w \, \partial_x u}{1 + u^2} \, dx$$

- **Program code:**

```
Int[v_*ArcTan[u_],x_Symbol] :=
  Module[{w=Block[{ShowSteps=False,StepCounter=Null}, Int[v,x]]},
  w*ArcTan[u] -
  Int[Regularize[w*D[u,x]/(1+u^2),x],x] /;
 InverseFunctionFreeQ[w,x]] /;
InverseFunctionFreeQ[u,x] &&
    Not[MatchQ[v, x^m_. /; FreeQ[m,x]]] &&
    FalseQ[FunctionOfLinear[v*ArcTan[u],x]]
```