$$\int \mathrm{Log}\left[c\,\left(a+b\,(d+e\,x)^n\right)^p\right]\,dx$$

- **Reference: G&R 2.728.1, CRC 499, A&S 4.1.49**
- **Derivation: Integration by parts**
- **Rule:**

$$\int \mathrm{Log}\left[c\,\left(b\,(d+e\,x)^n\right)^p\right]\,dx \longrightarrow \frac{(d+e\,x)\,\mathrm{Log}\left[c\,\left(b\,(d+e\,x)^n\right)^p\right]}{e}-n\,p\,x$$

- **Program code:**

```
Int[Log[c_.*(b_.*(d_.+e_.*x_)^n_.)^p_.],x_Symbol] :=
  (d+e*x)*Log[c*(b*(d+e*x)^n)^p]/e - n*p*x /;
FreeQ[{b,c,d,e,n,p},x]
```

- **Reference: G&R 2.728.1**
- **Derivation: Integration by parts**
- **Rule: If $n<0$, then**

$$\int \mathrm{Log}\left[c\,\left(a+b\,(d+e\,x)^n\right)^p\right]\,dx \longrightarrow \frac{(d+e\,x)\,\mathrm{Log}\left[c\,\left(a+b\,(d+e\,x)^n\right)^p\right]}{e}-b\,n\,p\int\frac{1}{b+a\,(d+e\,x)^{-n}}\,dx$$

- **Program code:**

```
Int[Log[c_.*(a_+b_.*(d_.+e_.*x_)^n_)^p_.],x_Symbol] :=
  (d+e*x)*Log[c*(a+b*(d+e*x)^n)^p]/e -
  Dist[b*n*p,Int[1/(b+a*(d+e*x)^(-n)),x]] /;
FreeQ[{a,b,c,d,e,p},x] && RationalQ[n] && n<0
```

- **Reference: G&R 2.728.1**
- **Derivation: Integration by parts**
- **Rule: If $\neg\,(n<0)$, then**

$$\int \mathrm{Log}\left[c\,\left(a+b\,(d+e\,x)^n\right)^p\right]\,dx \longrightarrow \frac{(d+e\,x)\,\mathrm{Log}\left[c\,\left(a+b\,(d+e\,x)^n\right)^p\right]}{e}-n\,p\,x+a\,n\,p\int\frac{1}{a+b\,(d+e\,x)^n}\,dx$$

- **Program code:**

```
Int[Log[c_.*(a_+b_.*(d_.+e_.*x_)^n_.)^p_.],x_Symbol] :=
  (d+e*x)*Log[c*(a+b*(d+e*x)^n)^p]/e - n*p*x +
  Dist[a*n*p,Int[1/(a+b*(d+e*x)^n),x]] /;
FreeQ[{a,b,c,d,e,n,p},x] && Not[RationalQ[n] && n<0]
```

$$\int (a + b \operatorname{Log}[c\ (d + e\ x)^n])^p \, dx$$

- **Reference: CRC 492**
- **Derivation: Primitive rule**
- **Basis:** $\partial_x \text{LogIntegral}[x] = \frac{1}{\text{Log}[x]}$
- **Rule:**

$$\int \frac{1}{\text{Log}[c\ (d + e\ x)]} \, dx \longrightarrow \frac{\text{LogIntegral}[c\ (d + e\ x)]}{c\ e}$$

- **Program code:**

```
Int[1/Log[c_.*(d_.+e_.*x_)],x_Symbol] :=
  LogIntegral[c*(d+e*x)]/(c*e) /;
FreeQ[{c,d,e},x]
```

- **Derivation: Primitive rule**
- **Basis:** $\partial_x \text{ExpIntegralEi}[x] = \frac{e^x}{x}$
- **Rule:**

$$\int \frac{1}{a + b\ \text{Log}[c\ (d + e\ x)^n]} \, dx \longrightarrow \frac{(d + e\ x)\ \text{ExpIntegralEi}\left[\frac{a + b\ \text{Log}[c\ (d + e\ x)^n]}{b\ n}\right]}{b\ e\ n\ e^{\frac{a}{b\ n}}\ (c\ (d + e\ x)^n)^{1/n}}$$

- **Program code:**

```
Int[1/(a_.+b_.*Log[c_.*(d_.+e_.*x_)^n_.]),x_Symbol] :=
  (d+e*x)*ExpIntegralEi[(a+b*Log[c*(d+e*x)^n])/(b*n)]/(b*e*n*E^(a/(b*n))*(c*(d+e*x)^n)^(1/n)) /;
FreeQ[{a,b,c,d,e,n},x]
```

- **Rule: If $b\,n > 0$, then**

$$\int \frac{1}{\sqrt{a + b\,\text{Log}[c\,(d + e\,x)^n]}}\,dx \longrightarrow \frac{\sqrt{\pi}\,\sqrt{b\,n}\,(d + e\,x)\,\text{Erfi}\left[\frac{\sqrt{a+b\,\text{Log}[c\,(d+e\,x)^n]}}{\sqrt{b\,n}}\right]}{b\,e\,n\,e^{\frac{a}{b\,n}}\,(c\,(d + e\,x)^n)^{1/n}}$$

- **Program code:**

```
Int[1/Sqrt[a_.+b_.*Log[c_.*(d_.+e_.*x_)^n_.]],x_Symbol] :=
  Sqrt[Pi]*Rt[b*n,2]*(d+e*x)*Erfi[Sqrt[a+b*Log[c*(d+e*x)^n]]/Rt[b*n,2]]/
    (b*e*n*E^(a/(b*n))*(c*(d+e*x)^n)^(1/n)) /;
FreeQ[{a,b,c,d,e,n},x] && PosQ[b*n]
```

- **Rule: If $\neg\,(b\,n > 0)$, then**

$$\int \frac{1}{\sqrt{a + b\,\text{Log}[c\,(d + e\,x)^n]}}\,dx \longrightarrow \frac{\sqrt{\pi}\,\sqrt{-b\,n}\,(d + e\,x)\,\text{Erf}\left[\frac{\sqrt{a+b\,\text{Log}[c\,(d+e\,x)^n]}}{\sqrt{-b\,n}}\right]}{b\,e\,n\,e^{\frac{a}{b\,n}}\,(c\,(d + e\,x)^n)^{1/n}}$$

- **Program code:**

```
Int[1/Sqrt[a_.+b_.*Log[c_.*(d_.+e_.*x_)^n_.]],x_Symbol] :=
  Sqrt[Pi]*Rt[-b*n,2]*(d+e*x)*Erf[Sqrt[a+b*Log[c*(d+e*x)^n]]/Rt[-b*n,2]]/
    (b*e*n*E^(a/(b*n))*(c*(d+e*x)^n)^(1/n)) /;
FreeQ[{a,b,c,d,e,n},x] && NegQ[b*n]
```

- **Reference: G&R 2.711.1, CRC 490**
- **Derivation: Integration by parts**
- **Rule: If $p > 0$, then**

$$\int (a + b\,\text{Log}[c\,(d + e\,x)^n])^p\,dx \longrightarrow \frac{(d + e\,x)\,(a + b\,\text{Log}[c\,(d + e\,x)^n])^p}{e} - b\,n\,p\int (a + b\,\text{Log}[c\,(d + e\,x)^n])^{p-1}\,dx$$

- **Program code:**

```
Int[(a_.+b_.*Log[c_.*(d_.+e_.*x_)^n_.])^p_,x_Symbol] :=
  (d+e*x)*(a+b*Log[c*(d+e*x)^n])^p/e -
  Dist[b*n*p,Int[(a+b*Log[c*(d+e*x)^n])^(p-1),x]] /;
FreeQ[{a,b,c,d,e,n},x] && RationalQ[p] && p>0
```

- **Derivation: Inverted integration by parts**
- **Rule: If $p < -1$, then**

$$\int (a + b \operatorname{Log}[c (d + e x)^n])^p \, dx \longrightarrow \frac{(d + e x) (a + b \operatorname{Log}[c (d + e x)^n])^{p+1}}{b e n (p + 1)} - \frac{1}{b n (p + 1)} \int (a + b \operatorname{Log}[c (d + e x)^n])^{p+1} \, dx$$

- **Program code:**

```
Int[(a_.+b_.*Log[c_.*(d_.+e_.*x_)^n_.])^p_,x_Symbol] :=
  (d+e*x)*(a+b*Log[c*(d+e*x)^n])^(p+1)/(b*e*n*(p+1)) -
  Dist[1/(b*n*(p+1)),Int[(a+b*Log[c*(d+e*x)^n])^(p+1),x]] /;
FreeQ[{a,b,c,d,e,n},x] && RationalQ[p] && p<-1
```

- **Derivation: Inverted integration by parts**
- **Rule: If $2 p \notin \mathbb{Z}$, then**

$$\int (a + b \operatorname{Log}[c (d + e x)^n])^p \, dx \longrightarrow \frac{(d + e x) \operatorname{Gamma}\left[p + 1, -\frac{a+b \operatorname{Log}[c (d+e x)^n]}{b n}\right] (a + b \operatorname{Log}[c (d + e x)^n])^p}{e \left(-\frac{a+b \operatorname{Log}[c (d+e x)^n]}{b n}\right)^p e^{\frac{a}{b n}} (c (d + e x)^n)^{1/n}}$$

- **Program code:**

```
Int[(a_.+b_.*Log[c_.*(d_.+e_.*x_)^n_.])^p_,x_Symbol] :=
  (d+e*x)*Gamma[p+1,-(a+b*Log[c*(d+e*x)^n])/(b*n)]*(a+b*Log[c*(d+e*x)^n])^p/
    (e*(-(a+b*Log[c*(d+e*x)^n])/(b*n))^p*E^(a/(b*n))*(c*(d+e*x)^n)^(1/n)) /;
FreeQ[{a,b,c,d,e,n,p},x] && Not[IntegerQ[2*p]]
```

$\int x^m \text{Log}[c\,(a + b\,x^n)^p]\,dx$

- **Reference: G&R 2.728.2**
- **Derivation: Primitive rule**
- **Basis:** $\partial_x \text{PolyLog}[2, -x] = -\frac{\text{Log}[1+x]}{x}$
- **Rule:**

$$\int \frac{\text{Log}[1 + b\,x^n]}{x}\,dx \longrightarrow -\frac{\text{PolyLog}[2, -b\,x^n]}{n}$$

- **Program code:**

```
Int[Log[1+b_.*x_^n_.]/x_,x_Symbol] :=
  -PolyLog[2,-b*x^n]/n /;
FreeQ[{b,n},x]
```

- **Derivation: Derivation: Algebraic expansion**
- **Basis: If** $a > 0$, $\text{Log}[a\,z] = \text{Log}[a] + \text{Log}[z]$
- **Rule: If** $a\,c > 0$, **then**

$$\int \frac{\text{Log}[c\,(a + b\,x^n)]}{x}\,dx \longrightarrow \text{Log}[a\,c]\,\text{Log}[x] + \int \frac{\text{Log}\left[1 + \frac{b\,x^n}{a}\right]}{x}\,dx$$

- **Program code:**

```
Int[Log[c_.*(a_+b_.*x_^n_.)]/x_,x_Symbol] :=
  Log[a*c]*Log[x] +
  Int[Log[1+b*x^n/a]/x,x] /;
FreeQ[{a,b,c,n},x] && PositiveQ[a*c]
```

- **Derivation: Integration by parts**
- **Rule:**

$$\int \frac{\mathrm{Log}[c\,(a+b\,x^n)^p]}{x}\,dx \longrightarrow \frac{\mathrm{Log}[c\,(a+b\,x^n)^p]\,\mathrm{Log}\left[-\frac{b\,x^n}{a}\right]}{n}-b\,p\int\frac{x^{n-1}\,\mathrm{Log}\left[-\frac{b\,x^n}{a}\right]}{a+b\,x^n}\,dx$$

- **Program code:**

```
Int[Log[c_.*(a_+b_.*x_^n_.)^p_.]/x_,x_Symbol] :=
  Log[c*(a+b*x^n)^p]*Log[-b*x^n/a]/n -
  Dist[b*p,Int[x^(n-1)*Log[-b*x^n/a]/(a+b*x^n),x]] /;
FreeQ[{a,b,c,n,p},x]
```

- **Reference: G&R 2.728.1, CRC 501, A&S 4.1.50'**
- **Rule: If** $m+1\neq 0$, **then**

$$\int x^m\,\mathrm{Log}[c\,(b\,x^n)^p]\,dx \longrightarrow \frac{x^{m+1}\,\mathrm{Log}[c\,(b\,x^n)^p]}{m+1}-\frac{n\,p\,x^{m+1}}{(m+1)^2}$$

- **Program code:**

```
Int[x_^m_.*Log[c_.*(b_.*x_^n_.)^p_.],x_Symbol] :=
  x^(m+1)*Log[c*(b*x^n)^p]/(m+1) - n*p*x^(m+1)/(m+1)^2 /;
FreeQ[{b,c,m,n,p},x] && NonzeroQ[m+1]
```

- **Reference: G&R 2.728.1, CRC 501, A&S 4.1.50'**
- **Rule: If** $m+1\neq 0\ \wedge\ m-n+1\neq 0$, **then**

$$\int x^m\,\mathrm{Log}[c\,(a+b\,x^n)^p]\,dx \longrightarrow \frac{x^{m+1}\,\mathrm{Log}[c\,(a+b\,x^n)^p]}{m+1}-\frac{b\,n\,p}{m+1}\int\frac{x^{m+n}}{a+b\,x^n}\,dx$$

- **Program code:**

```
Int[x_^m_.*Log[c_.*(a_+b_.*x_^n_.)^p_.],x_Symbol] :=
  x^(m+1)*Log[c*(a+b*x^n)^p]/(m+1) -
  Dist[b*n*p/(m+1),Int[x^(m+n)/(a+b*x^n),x]] /;
FreeQ[{a,b,c,m,n,p},x] && NonzeroQ[m+1] && NonzeroQ[m-n+1]
```

$$\int x^m \left(a+b\,\text{Log}[c\,x^n]\right)^p\,dx$$

- **Rule: If** $m+1\neq 0$**, then**

$$\int \frac{x^m}{a+b\,\text{Log}[c\,x^n]}\,dx \longrightarrow \frac{x^{m+1}\,\text{ExpIntegralEi}\left[\frac{(m+1)\,(a+b\,\text{Log}[c\,x^n])}{b\,n}\right]}{b\,n\,e^{\frac{a\,(m+1)}{b\,n}}\,(c\,x^n)^{\frac{m+1}{n}}}$$

- **Program code:**

```
Int[x_^m_./(a_.+b_.*Log[c_.*x_^n_.]),x_Symbol] :=
  x^(m+1)*ExpIntegralEi[(m+1)*(a+b*Log[c*x^n])/(b*n)]/(b*n*E^(a*(m+1)/(b*n))*(c*x^n)^((m+1)/n)) /;
FreeQ[{a,b,c,m,n},x] && NonzeroQ[m+1]
```

- **Rule: If** $m+1\neq 0 \bigwedge \frac{m+1}{b\,n}>0$**, then**

$$\int \frac{x^m}{\sqrt{a+b\,\text{Log}[c\,x^n]}}\,dx \longrightarrow \frac{\sqrt{\pi}\,x^{m+1}\,\text{Erfi}\left[\sqrt{\frac{m+1}{b\,n}}\,\sqrt{a+b\,\text{Log}[c\,x^n]}\right]}{b\,n\sqrt{\frac{m+1}{b\,n}}\,e^{\frac{a\,(m+1)}{b\,n}}\,(c\,x^n)^{\frac{m+1}{n}}}$$

- **Program code:**

```
Int[x_^m_./Sqrt[a_.+b_.*Log[c_.*x_^n_.]],x_Symbol] :=
  Sqrt[Pi]*x^(m+1)*Erfi[Rt[(m+1)/(b*n),2]*Sqrt[a+b*Log[c*x^n]]]/
    (b*n*Rt[(m+1)/(b*n),2]*E^(a*(m+1)/(b*n))*(c*x^n)^((m+1)/n)) /;
FreeQ[{a,b,c,m,n},x] && NonzeroQ[m+1] && PosQ[(m+1)/(b*n)]
```

- **Rule: If** $m+1\neq 0 \bigwedge \neg\left(\frac{m+1}{b\,n}>0\right)$**, then**

$$\int \frac{x^m}{\sqrt{a+b\,\text{Log}[c\,x^n]}}\,dx \longrightarrow \frac{\sqrt{\pi}\,x^{m+1}\,\text{Erf}\left[\sqrt{-\frac{m+1}{b\,n}}\,\sqrt{a+b\,\text{Log}[c\,x^n]}\right]}{b\,n\sqrt{-\frac{m+1}{b\,n}}\,e^{\frac{a\,(m+1)}{b\,n}}\,(c\,x^n)^{\frac{m+1}{n}}}$$

- **Program code:**

```
Int[x_^m_./Sqrt[a_.+b_.*Log[c_.*x_^n_.]],x_Symbol] :=
  Sqrt[Pi]*x^(m+1)*Erf[Rt[-(m+1)/(b*n),2]*Sqrt[a+b*Log[c*x^n]]]/
    (b*n*Rt[-(m+1)/(b*n),2]*E^(a*(m+1)/(b*n))*(c*x^n)^((m+1)/n)) /;
FreeQ[{a,b,c,m,n},x] && NonzeroQ[m+1] && NegQ[(m+1)/(b*n)]
```

- **Reference: G&R 2.721.1, CRC 496, A&S 4.1.51**
- **Derivation: Integration by parts**
- **Rule: If $m+1 \neq 0 \land p > 0$, then**

$$\int x^m (a+b\,\mathrm{Log}[c\,x^n])^p\,dx \longrightarrow \frac{x^{m+1}(a+b\,\mathrm{Log}[c\,x^n])^p}{m+1} - \frac{b\,n\,p}{m+1}\int x^m (a+b\,\mathrm{Log}[c\,x^n])^{p-1}\,dx$$

- **Program code:**

```
Int[x_^m_.*(a_.+b_.*Log[c_.*x_^n_.])^p_,x_Symbol] :=
  x^(m+1)*(a+b*Log[c*x^n])^p/(m+1) -
  Dist[b*n*p/(m+1),Int[x^m*(a+b*Log[c*x^n])^(p-1),x]] /;
FreeQ[{a,b,c,m,n},x] && NonzeroQ[m+1] && RationalQ[p] && p>0
```

- **Reference: G&R 2.724.1, CRC 495**
- **Derivation: Inverted integration by parts**
- **Rule: If $m+1 \neq 0 \land p < -1$, then**

$$\int x^m (a+b\,\mathrm{Log}[c\,x^n])^p\,dx \longrightarrow \frac{x^{m+1}(a+b\,\mathrm{Log}[c\,x^n])^p}{m+1} - \frac{b\,n\,p}{m+1}\int x^m (a+b\,\mathrm{Log}[c\,x^n])^{p-1}\,dx$$

- **Program code:**

```
Int[x_^m_.*(a_.+b_.*Log[c_.*x_^n_.])^p_,x_Symbol] :=
  x^(m+1)*(a+b*Log[c*x^n])^(p+1)/(b*n*(p+1)) -
  Dist[(m+1)/(b*n*(p+1)),Int[x^m*(a+b*Log[c*x^n])^(p+1),x]] /;
FreeQ[{a,b,c,m,n},x] && NonzeroQ[m+1] && RationalQ[p] && p<-1
```

- **Rule: If $m+1 \neq 0$, then**

$$\int x^m (a+b\,\mathrm{Log}[c\,x^n])^p\,dx \longrightarrow \frac{x^{m+1}\,\mathrm{Gamma}\left[p+1,-\frac{(m+1)(a+b\,\mathrm{Log}[c\,x^n])}{b\,n}\right](a+b\,\mathrm{Log}[c\,x^n])^p}{(m+1)\,e^{\frac{a(m+1)}{b\,n}}\,(c\,x^n)^{\frac{m+1}{n}}\left(-\frac{(m+1)(a+b\,\mathrm{Log}[c\,x^n])}{b\,n}\right)^p}$$

- **Program code:**

```
Int[x_^m_.*(a_.+b_.*Log[c_.*x_^n_.])^p_,x_Symbol] :=
  x^(m+1)*Gamma[p+1,-(m+1)*(a+b*Log[c*x^n])/(b*n)]*(a+b*Log[c*x^n])^p/
    ((m+1)*E^(a*(m+1)/(b*n))*(c*x^n)^((m+1)/n)*(-(m+1)*(a+b*Log[c*x^n])/(b*n))^p) /;
FreeQ[{a,b,c,m,n,p},x] && NonzeroQ[m+1]
```

- **Note: Need a rule for arbitrarily deep nesting of powers!**

- **Rule:**

$$\int \mathrm{Log}[a\,(b\,x^n)^p]^q\,dx \longrightarrow \mathrm{Subst}\left[\int \mathrm{Log}[x^{n\,p}]^q\,dx,\ x^{n\,p},\ a\,(b\,x^n)^p\right]$$

- **Program code:**

```
Int[Log[a_.*(b_.*x_^n_.)^p_]^q_.,x_Symbol] :=
  Subst[Int[Log[x^(n*p)]^q,x],x^(n*p),a*(b*x^n)^p] /;
FreeQ[{a,b,n,p,q},x]
```

- **Rule:**

$$\int \mathrm{Log}[a\,(b\,(c\,x^n)^p)^q]^r\,dx \longrightarrow \mathrm{Subst}\left[\int \mathrm{Log}[x^{n\,p\,q}]^r\,dx,\ x^{n\,p\,q},\ a\,(b\,(c\,x^n)^p)^q\right]$$

- **Program code:**

```
Int[Log[a_.*(b_.*(c_.*x_^n_.)^p_)^q_]^r_.,x_Symbol] :=
  Subst[Int[Log[x^(n*p*q)]^r,x],x^(n*p*q),a*(b*(c*x^n)^p)^q] /;
FreeQ[{a,b,c,n,p,q,r},x]
```

- **Rule: If $m+1\neq 0$, then**

$$\int x^m\,\mathrm{Log}[a\,(b\,x^n)^p]^q\,dx \longrightarrow \mathrm{Subst}\left[\int x^m\,\mathrm{Log}[x^{n\,p}]^q\,dx,\ x^{n\,p},\ a\,(b\,x^n)^p\right]$$

- **Program code:**

```
Int[x_^m_.*Log[a_.*(b_.*x_^n_.)^p_]^q_.,x_Symbol] :=
  Subst[Int[x^m*Log[x^(n*p)]^q,x],x^(n*p),a*(b*x^n)^p] /;
FreeQ[{a,b,m,n,p,q},x] && NonzeroQ[m+1] && Not[x^(n*p)===a*(b*x^n)^p]
```

- **Rule: If $m+1\neq 0$, then**

$$\int x^m\,\mathrm{Log}[a\,(b\,(c\,x^n)^p)^q]^r\,dx \longrightarrow \mathrm{Subst}\left[\int x^m\,\mathrm{Log}[x^{n\,p\,q}]^r\,dx,\ x^{n\,p\,q},\ a\,(b\,(c\,x^n)^p)^q\right]$$

- **Program code:**

```
Int[x_^m_.*Log[a_.*(b_.*(c_.*x_^n_.)^p_)^q_]^r_.,x_Symbol] :=
  Subst[Int[x^m*Log[x^(n*p*q)]^r,x],x^(n*p*q),a*(b*(c*x^n)^p)^q] /;
FreeQ[{a,b,c,m,n,p,q,r},x] && NonzeroQ[m+1] && Not[x^(n*p*q)===a*(b*(c*x^n)^p)^q]
```

$$\int x^m \operatorname{Log}[c (a+b x)^n]^p \, dx$$

- **Rule: If** $m > 0 \land p > 0$**, then**
- **Rule:**

$$\int x^m \operatorname{Log}[c (a+b x)^n]^p \, dx \longrightarrow \frac{x^m (a+b x) \operatorname{Log}[c (a+b x)^n]^p}{b (m+1)} - \frac{a m}{b (m+1)} \int x^{m-1} \operatorname{Log}[c (a+b x)^n]^p \, dx - \frac{n p}{m+1} \int x^m \operatorname{Log}[c (a+b x)^n]^{p-1} \, dx$$

- **Program code:**

```
Int[x_^m_.*Log[c_.*(a_+b_.*x_)^n_.]^p_,x_Symbol] :=
  x^m*(a+b*x)*Log[c*(a+b*x)^n]^p/(b*(m+1)) -
  Dist[a*m/(b*(m+1)),Int[x^(m-1)*Log[c*(a+b*x)^n]^p,x]] -
  Dist[n*p/(m+1),Int[x^m*Log[c*(a+b*x)^n]^(p-1),x]] /;
FreeQ[{a,b,c,n},x] && RationalQ[{m,p}] && m>0 && p>0
```

- **Rule: If** $p > 0$**, then**

$$\int \frac{\operatorname{Log}[c (a+b x)^n]^p}{x^2} \, dx \longrightarrow -\frac{(a+b x) \operatorname{Log}[c (a+b x)^n]^p}{a x} + \frac{b n p}{a} \int \frac{\operatorname{Log}[c (a+b x)^n]^{p-1}}{x} \, dx$$

- **Program code:**

```
Int[Log[c_.*(a_+b_.*x_)^n_.]^p_/x_^2,x_Symbol] :=
  -(a+b*x)*Log[c*(a+b*x)^n]^p/(a*x) +
  Dist[b*n*p/a,Int[Log[c*(a+b*x)^n]^(p-1)/x,x]] /;
FreeQ[{a,b,c,n},x] && RationalQ[p] && p>0
```

- **Rule: If** $m < -1 \land m \neq -2 \land p > 0$**, then**

$$\int x^m \operatorname{Log}[c (a+b x)^n]^p \, dx \longrightarrow \frac{x^{m+1} (a+b x) \operatorname{Log}[c (a+b x)^n]^p}{a (m+1)} - \frac{b (m+2)}{a (m+1)} \int x^{m+1} \operatorname{Log}[c (a+b x)^n]^p \, dx - \frac{b n p}{a (m+1)} \int x^{m+1} \operatorname{Log}[c (a+b x)^n]^{p-1} \, dx$$

- **Program code:**

```
Int[x_^m_.*Log[c_.*(a_+b_.*x_)^n_.]^p_,x_Symbol] :=
  x^(m+1)*(a+b*x)*Log[c*(a+b*x)^n]^p/(a*(m+1)) -
  Dist[(b*(m+2))/(a*(m+1)),Int[x^(m+1)*Log[c*(a+b*x)^n]^p,x]] -
  Dist[b*n*p/(a*(m+1)),Int[x^(m+1)*Log[c*(a+b*x)^n]^(p-1),x]] /;
FreeQ[{a,b,c,n},x] && RationalQ[{m,p}] && m<-1 && m!=-2 && p>0
```

- **Rule: If** $m \in \mathbb{Z} \wedge m > 0 \wedge \neg (p \in \mathbb{Q} \wedge p > 0)$**, then**

$$\int x^m \operatorname{Log}[c\,(a+b\,x)^n]^p \, dx \longrightarrow \frac{1}{b}\,\operatorname{Subst}\left[\int \left(-\frac{a}{b}+\frac{x}{b}\right)^m \operatorname{Log}[c\,x^n]^p \, dx,\ x,\ a+b\,x\right]$$

- **Program code:**

```
Int[x_^m_.*Log[c_.*(a_+b_.*x_)^n_.]^p_,x_Symbol] :=
  Dist[1/b,Subst[Int[(-a/b+x/b)^m*Log[c*x^n]^p,x],x,a+b*x]] /;
FreeQ[{a,b,c,n,p},x] && IntegerQ[m] && m>0 && Not[RationalQ[p] && p>0]
```

$$\int \frac{\text{Log}[c\,(a+b\,x)^n]^p}{(d+e\,x)^m}\,dx$$

- **Reference: G&R 2.728.2**
- **Derivation: Primitive rule**
- **Basis:** $\partial_x\left(-\frac{\text{PolyLog}[2,d+e\,x]}{e}\right)=\frac{\text{Log}[1-d-e\,x]}{d+e\,x}$
- **Rule: If** $a\,c\,e-b\,c\,d-e=0$**, then**

$$\int \frac{\text{Log}[c\,(a+b\,x)]}{d+e\,x}\,dx \longrightarrow -\frac{\text{PolyLog}[2,\,1-a\,c-b\,c\,x]}{e}$$

- **Program code:**

```
Int[Log[c_.*(a_.+b_.*x_)]/(d_+e_.*x_),x_Symbol] :=
  -PolyLog[2,1-a*c-b*c*x]/e /;
FreeQ[{a,b,c,d,e},x] && ZeroQ[a*c*e-b*c*d-e]
```

- **Derivation: Integration by parts**
- **Rule: If** $b\,d-a\,e\neq 0$**, then**

$$\int \frac{\text{Log}[c\,(a+b\,x)^n]}{d+e\,x}\,dx \longrightarrow \frac{1}{e}\,\text{Log}[c\,(a+b\,x)^n]\,\text{Log}\left[\frac{b\,(d+e\,x)}{b\,d-a\,e}\right]+\frac{n}{e}\,\text{PolyLog}\left[2,\,-\frac{e\,(a+b\,x)}{b\,d-a\,e}\right]$$

- **Program code:**

```
Int[Log[c_.*(a_.+b_.*x_)^n_.]/(d_+e_.*x_),x_Symbol] :=
  Log[c*(a+b*x)^n]*Log[b*(d+e*x)/(b*d-a*e)]/e +
  n*PolyLog[2,-e*(a+b*x)/(b*d-a*e)]/e /;
FreeQ[{a,b,c,d,e,n},x] && NonzeroQ[b*d-a*e]
```

- **Derivation: Integration by parts**
- **Rule: If** $p > 0 \wedge b\,d - a\,e \neq 0$, **then**

$$\int \frac{\text{Log}[c\,(a+b\,x)^n]^p}{d+e\,x}\,dx \longrightarrow \frac{1}{e}\,\text{Log}[c\,(a+b\,x)^n]^p\,\text{Log}\Big[\frac{b\,(d+e\,x)}{b\,d-a\,e}\Big] - \frac{b\,n\,p}{e}\int \frac{\text{Log}[c\,(a+b\,x)^n]^{p-1}\,\text{Log}\Big[\frac{b\,(d+e\,x)}{b\,d-a\,e}\Big]}{a+b\,x}\,dx$$

- **Program code:**

```
Int[Log[c_.*(a_+b_.*x_)^n_.]^p_./(d_.+e_.*x_),x_Symbol] :=
  Log[c*(a+b*x)^n]^p*Log[b*(d+e*x)/(b*d-a*e)]/e -
  Dist[b*n*p/e,Int[Log[c*(a+b*x)^n]^(p-1)*Log[b*(d+e*x)/(b*d-a*e)]/(a+b*x),x]] /;
FreeQ[{a,b,c,d,e,n},x] && RationalQ[p] && p>0 && NonzeroQ[b*d-a*e]
```

- **Note:** $\text{Log}[z] = -\text{PolyLog}[1, 1-z]$
- **Rule: If** $p > 0 \wedge b\,d - a\,e = 0 \wedge a\,g\,h - b\,(f\,h-1) = 0$, **then**

$$\int \frac{\text{Log}[c\,(a+b\,x)^n]^p\,\text{Log}[h\,(f+g\,x)]}{d+e\,x}\,dx \longrightarrow -\frac{\text{Log}[c\,(a+b\,x)^n]^p\,\text{PolyLog}[2,\,1-h\,(f+g\,x)]}{e} + \frac{b\,n\,p}{e}\int \frac{\text{Log}[c\,(a+b\,x)^n]^{p-1}\,\text{PolyLog}[2,\,1-h\,(f+g\,x)]}{a+b\,x}\,dx$$

- **Program code:**

```
Int[Log[c_.*(a_+b_.*x_)^n_.]^p_.*Log[h_.*(f_.+g_.*x_)]/(d_+e_.*x_),x_Symbol] :=
  Module[{q=Simplify[1-h*(f+g*x)]},
  -Log[c*(a+b*x)^n]^p*PolyLog[2,q]/e +
  Dist[b*n*p/e,Int[Log[c*(a+b*x)^n]^(p-1)*PolyLog[2,q]/(a+b*x),x]]] /;
FreeQ[{a,b,c,d,e,f,g,h,n},x] && RationalQ[p] && p>0 && ZeroQ[b*d-a*e] && ZeroQ[a*g*h-b*(f*h-1)]
```

- **Rule: If** $p > 0 \wedge b\,d - a\,e = 0 \wedge a\,g - b\,f = 0$, **then**

$$\int \frac{\text{Log}[c\,(a+b\,x)^n]^p\,\text{PolyLog}[m,\,h\,(f+g\,x)]}{d+e\,x}\,dx \longrightarrow \frac{\text{Log}[c\,(a+b\,x)^n]^p\,\text{PolyLog}[m+1,\,h\,(f+g\,x)]}{e} - \frac{b\,n\,p}{e}\int \frac{\text{Log}[c\,(a+b\,x)^n]^{p-1}\,\text{PolyLog}[m+1,\,h\,(f+g\,x)]}{a+b\,x}\,dx$$

- **Program code:**

```
Int[Log[c_.*(a_+b_.*x_)^n_.]^p_.*PolyLog[m_,h_.*(f_.+g_.*x_)]/(d_+e_.*x_),x_Symbol] :=
  Log[c*(a+b*x)^n]^p*PolyLog[m+1,h*(f+g*x)]/e -
  Dist[b*n*p/e,Int[Log[c*(a+b*x)^n]^(p-1)*PolyLog[m+1,h*(f+g*x)]/(a+b*x),x]] /;
FreeQ[{a,b,c,d,e,f,g,h,m,n},x] && RationalQ[p] && p>0 && ZeroQ[b*d-a*e] && ZeroQ[a*g-b*f]
```

- **Derivation: Integration by parts**
- **Rule: If $m, p \in \mathbb{Z} \land m < -1 \land p > 0$, then**
- **Rule:**

$$\int (d+e\,x)^m \operatorname{Log}\left[c\,(a+b\,x)^n\right]^p dx \longrightarrow$$
$$\frac{(d+e\,x)^{m+1} \operatorname{Log}\left[c\,(a+b\,x)^n\right]^p}{e\,(m+1)} - \frac{b\,n\,p}{e\,(m+1)} \int \frac{(d+e\,x)^{m+1} \operatorname{Log}\left[c\,(a+b\,x)^n\right]^{p-1}}{a+b\,x} dx$$

- **Program code:**

```
Int[(d_.+e_.*x_)^m_.*Log[c_.*(a_.+b_.*x_)^n_.]^p_,x_Symbol] :=
  (d+e*x)^(m+1)*Log[c*(a+b*x)^n]^p/(e*(m+1)) -
  Dist[b*n*p/(e*(m+1)),Int[Regularize[(d+e*x)^(m+1)*Log[c*(a+b*x)^n]^(p-1)/(a+b*x),x],x]] /;
FreeQ[{a,b,c,d,e,n},x] && IntegersQ[m,p] && m<-1 && p>0
```

$$\int \text{Log}[c\,(a+b\,x^m)^n]^p\,dx$$

- **Note: The b/x in the resulting integrand will be transformed to b x by the rule** $\int \frac{f[x^n]}{x}\,dx \to \frac{\text{Subst}\left[\int \frac{f[x]}{x}\,dx,\,x,\,x^n\right]}{n}$

- **Rule: If** $p \in \mathbb{Z} \land p > 0$**, then**

$$\int \text{Log}\left[c\left(a+\frac{b}{x}\right)^n\right]^p dx \to \frac{(b+a\,x)\,\text{Log}\left[c\left(a+\frac{b}{x}\right)^n\right]^p}{a}+\frac{b\,n\,p}{a}\int \frac{\text{Log}\left[c\left(a+\frac{b}{x}\right)^n\right]^{p-1}}{x}\,dx$$

- **Program code:**

```
Int[Log[c_.*(a_+b_./x_)^n_.]^p_, x_Symbol] :=
  (b+a*x)*Log[c*(a+b/x)^n]^p/a +
  Dist[b/a*n*p,Int[Log[c*(a+b/x)^n]^(p-1)/x,x]] /;
FreeQ[{a,b,c,n},x] && IntegerQ[p] && p>0
```

- **Rule:**

$$\int \mathrm{Log}\left[c\,\left(a+b\,x^2\right)^n\right]^2\,dx \longrightarrow x\,\mathrm{Log}\left[c\,\left(a+b\,x^2\right)^n\right]^2+8\,n^2\,x-4\,n\,x\,\mathrm{Log}\left[c\,\left(a+b\,x^2\right)^n\right]+\frac{1}{\sqrt{-b}}\left(n\,\sqrt{a}\right)$$
$$\left(4\,n\,\mathrm{Log}\left[\frac{-\sqrt{a}+\sqrt{-b}\,x}{\sqrt{a}+\sqrt{-b}\,x}\right]-4\,n\,\mathrm{ArcTanh}\left[\frac{\sqrt{-b}\,x}{\sqrt{a}}\right]\left(\mathrm{Log}\left[-\frac{\sqrt{a}}{\sqrt{-b}}+x\right]+\mathrm{Log}\left[\frac{\sqrt{a}}{\sqrt{-b}}+x\right]\right)-\right.$$
$$n\,\mathrm{Log}\left[-\frac{\sqrt{a}}{\sqrt{-b}}+x\right]^2+n\,\mathrm{Log}\left[\frac{\sqrt{a}}{\sqrt{-b}}+x\right]^2-2\,n\,\mathrm{Log}\left[\frac{\sqrt{a}}{\sqrt{-b}}+x\right]\mathrm{Log}\left[\frac{1}{2}-\frac{\sqrt{-b}\,x}{2\,\sqrt{a}}\right]+$$
$$2\,n\,\mathrm{Log}\left[-\frac{\sqrt{a}}{\sqrt{-b}}+x\right]\mathrm{Log}\left[\frac{1}{2}\left(1+\frac{\sqrt{-b}\,x}{\sqrt{a}}\right)\right]+4\,\mathrm{ArcTanh}\left[\frac{\sqrt{-b}\,x}{\sqrt{a}}\right]\mathrm{Log}\left[c\,\left(a+b\,x^2\right)^n\right]+$$
$$\left.2\,n\,\mathrm{PolyLog}\left[2,\,\frac{1}{2}-\frac{\sqrt{-b}\,x}{2\,\sqrt{a}}\right]-2\,n\,\mathrm{PolyLog}\left[2,\,\frac{1}{2}\left(1+\frac{\sqrt{-b}\,x}{\sqrt{a}}\right)\right]\right)$$

- **Program code:**

```
Int[Log[c_.*(a_+b_.*x_^2)^n_.]^2, x_Symbol] :=
  x*Log[c*(a+b*x^2)^n]^2 +
    8*n^2*x -
    4*n*x*Log[c*(a+b*x^2)^n] +
    (n*Sqrt[a]/Sqrt[-b])*(
      4*n*Log[(-Sqrt[a]+Sqrt[-b]*x)/(Sqrt[a]+Sqrt[-b]*x)] -
      4*n*ArcTanh[Sqrt[-b]*x/Sqrt[a]]*(Log[-Sqrt[a]/Sqrt[-b]+x] + Log[Sqrt[a]/Sqrt[-b]+x]) -
      n*Log[-Sqrt[a]/Sqrt[-b]+x]^2 +
      n*Log[Sqrt[a]/Sqrt[-b]+x]^2 -
      2*n*Log[Sqrt[a]/Sqrt[-b]+x]*Log[1/2-Sqrt[-b]*x/(2*Sqrt[a])] +
      2*n*Log[-Sqrt[a]/Sqrt[-b]+x]*Log[(1+Sqrt[-b]*x/Sqrt[a])/2] +
      4*ArcTanh[Sqrt[-b]*x/Sqrt[a]]*Log[c*(a+b*x^2)^n] +
      2*n*PolyLog[2,1/2-Sqrt[-b]*x/(2*Sqrt[a])] -
      2*n*PolyLog[2,(1+Sqrt[-b]*x/Sqrt[a])/2]) /;
FreeQ[{a,b,c,n},x]
```

$$\int \text{Log}\left[d \left(a + b x + c x^2\right)^n\right]^p dx$$

- **Derivation: Integration by parts**
- **Rule:**

$$\int \text{Log}[d (a + x (b + c x))^n]^2 dx \longrightarrow x \text{Log}\left[d \left(a + b x + c x^2\right)^n\right]^2 - 2 b n \int \frac{x \text{Log}\left[d \left(a + b x + c x^2\right)^n\right]}{a + b x + c x^2} dx - 4 c n \int \frac{x^2 \text{Log}\left[d \left(a + b x + c x^2\right)^n\right]}{a + b x + c x^2} dx$$

- **Program code:**

```
Int[Log[d_.*(a_.+b_.*x_+c_.*x_^2)^n_.]^2,x_Symbol] :=
  x*Log[d*(a+b*x+c*x^2)^n]^2 -
  Dist[2*b*n,Int[x*Log[d*(a+b*x+c*x^2)^n]/(a+b*x+c*x^2),x]] -
  Dist[4*c*n,Int[x^2*Log[d*(a+b*x+c*x^2)^n]/(a+b*x+c*x^2),x]] /;
FreeQ[{a,b,c,d,n},x]
```

$$\int x^m \,\text{Log}[a\,\text{Log}[b\,x^n]^p]\,dx$$

- **Derivation: Integration by parts**
- **Rule:**

$$\int \text{Log}[a\,\text{Log}[b\,x^n]^p]\,dx \longrightarrow x\,\text{Log}[a\,\text{Log}[b\,x^n]^p] - n\,p\int \frac{1}{\text{Log}[b\,x^n]}\,dx$$

- **Program code:**

```
Int[Log[a_.*Log[b_.*x_^n_.]^p_.],x_Symbol] :=
  x*Log[a*Log[b*x^n]^p] -
  Dist[n*p,Int[1/Log[b*x^n],x]] /;
FreeQ[{a,b,n,p},x]
```

- **Derivation: Integration by parts**
- **Rule:**

$$\int \frac{\text{Log}[a\,\text{Log}[b\,x^n]^p]}{x}\,dx \longrightarrow \frac{\text{Log}[b\,x^n]\,(-p+\text{Log}[a\,\text{Log}[b\,x^n]^p])}{n}$$

- **Program code:**

```
Int[Log[a_.*Log[b_.*x_^n_.]^p_.]/x_,x_Symbol] :=
  Log[b*x^n]*(-p+Log[a*Log[b*x^n]^p])/n /;
FreeQ[{a,b,n,p},x]
```

- **Derivation: Integration by parts**
- **Rule: If** $m+1 \neq 0$**, then**

$$\int x^m\,\text{Log}[a\,\text{Log}[b\,x^n]^p]\,dx \longrightarrow \frac{x^{m+1}\,\text{Log}[a\,\text{Log}[b\,x^n]^p]}{m+1} - \frac{n\,p}{m+1}\int \frac{x^m}{\text{Log}[b\,x^n]}\,dx$$

- **Program code:**

```
Int[x_^m_.*Log[a_.*Log[b_.*x_^n_.]^p_.],x_Symbol] :=
  x^(m+1)*Log[a*Log[b*x^n]^p]/(m+1) -
  Dist[n*p/(m+1),Int[x^m/Log[b*x^n],x]] /;
FreeQ[{a,b,m,n,p},x] && NonzeroQ[m+1]
```

$$\int \frac{\mathrm{Log}\left[\frac{a+b\,x}{c+d\,x}\right]^m}{x}\,dx$$

- **Basis:** $\frac{f\left[\frac{a+b\,x}{c+d\,x}\right]}{x} = \frac{f\left[\frac{a}{c}+\frac{(b\,c-a\,d)\,x}{c\,(c+d\,x)}\right]}{\frac{(b\,c-a\,d)\,x}{c+d\,x}}\,\partial_x\frac{(b\,c-a\,d)\,x}{c+d\,x} - \frac{f\left[\frac{b}{d}-\frac{b\,c-a\,d}{(c+d\,x)\,d}\right]}{\frac{b\,c-a\,d}{c+d\,x}}\,\partial_x\left(\frac{b\,c-a\,d}{c+d\,x}\right)$
- **Note: This linearizing substitution is valid for any function of** $\frac{a+b\,x}{c+d\,x}$ **over** x.
- **Rule: If** $m\in\mathbb{Z}\land m>0\land b\,c-a\,d\neq 0$, **then**

$$\int \frac{\mathrm{Log}\left[\frac{a+b\,x}{c+d\,x}\right]^m}{x}\,dx \to \mathrm{Subst}\left[\int\frac{\mathrm{Log}\left[\frac{a}{c}+\frac{x}{c}\right]^m}{x}\,dx,\,x,\,\frac{(b\,c-a\,d)\,x}{c+d\,x}\right]-\mathrm{Subst}\left[\int\frac{\mathrm{Log}\left[\frac{b}{d}-\frac{x}{d}\right]^m}{x}\,dx,\,x,\,\frac{b\,c-a\,d}{c+d\,x}\right]$$

- **Program code:**

```
Int[Log[(a_.+b_.*x_)/(c_+d_.*x_)]^m_./x_,x_Symbol] :=
  Subst[Int[Log[a/c+x/c]^m/x,x],x,(b*c-a*d)*x/(c+d*x)] -
  Subst[Int[Log[b/d-x/d]^m/x,x],x,(b*c-a*d)/(c+d*x)] /;
FreeQ[{a,b,c,d},x] && IntegerQ[m] && m>0 && NonzeroQ[b*c-a*d]
```

$$\int \frac{A + B \operatorname{Log}[c + d\,x]}{\sqrt{a + b \operatorname{Log}[c + d\,x]}} \, dx$$

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{A+B\,z}{\sqrt{a+b\,z}} = \frac{b\,A-a\,B}{b\,\sqrt{a+b\,z}} + \frac{B\,\sqrt{a+b\,z}}{b}$
- **Rule: If** $b\,A - a\,B \neq 0$**, then**

$$\int \frac{A + B \operatorname{Log}[c + d\,x]}{\sqrt{a + b \operatorname{Log}[c + d\,x]}} \, dx \longrightarrow \frac{b\,A - a\,B}{b} \int \frac{1}{\sqrt{a + b \operatorname{Log}[c + d\,x]}} \, dx + \frac{B}{b} \int \sqrt{a + b \operatorname{Log}[c + d\,x]} \, dx$$

- **Program code:**

```
Int[(A_.+B_.*Log[c_.+d_.*x_])/Sqrt[a_+b_.*Log[c_.+d_.*x_]],x_Symbol] :=
  Dist[(b*A-a*B)/b,Int[1/Sqrt[a+b*Log[c+d*x]],x]] +
  Dist[B/b,Int[Sqrt[a+b*Log[c+d*x]],x]] /;
FreeQ[{a,b,c,d,A,B},x] && NonzeroQ[b*A-a*B]
```

$$\int f^{a\,\mathrm{Log}[u]}\,dx$$

- **Derivation: Algebraic simplification**
- **Basis:** $f^{a\,\mathrm{Log}[g]} = g^{a\,\mathrm{Log}[f]}$
- **Rule:**

$$\int f^{a\,\mathrm{Log}[u]}\,dx \longrightarrow \int u^{a\,\mathrm{Log}[f]}\,dx$$

- **Program code:**

```
Int[f_^(a_.*Log[u_]),x_Symbol] :=
  Int[u^(a*Log[f]),x] /;
FreeQ[{a,f},x]
```

$$\int \frac{f[\text{Log}[a\,x^n]]}{x}\,dx$$

- **Derivation: Integration by substitution**
- **Basis:** $\frac{f[\text{Log}[a\,x^n]]}{x} = \frac{1}{n}\,f[\text{Log}[a\,x^n]]\,\partial_x \text{Log}[a\,x^n]$
- **Rule:**

$$\int \frac{f[\text{Log}[a\,x^n]]}{x}\,dx \longrightarrow \frac{1}{n}\,\text{Subst}\left[\int f[x]\,dx,\ x,\ \text{Log}[a\,x^n]\right]$$

- **Program code:**

```
If[ShowSteps,

Int[u_/x_,x_Symbol] :=
  Module[{lst=FunctionOfLog[u,x]},
  ShowStep["","Int[f[Log[a*x^n]]/x,x]","Subst[Int[f[x],x],x,Log[a*x^n]]/n",Hold[
  Dist[1/lst[[3]],Subst[Int[lst[[1]],x],x,Log[lst[[2]]]]]]] /;
 Not[FalseQ[lst]]] /;
SimplifyFlag && NonsumQ[u],

Int[u_/x_,x_Symbol] :=
  Module[{lst=FunctionOfLog[u,x]},
  Dist[1/lst[[3]],Subst[Int[lst[[1]],x],x,Log[lst[[2]]]]] /;
 Not[FalseQ[lst]]] /;
NonsumQ[u]]
```

- **Derivation: Algebraic simplification**
- **Basis:** $\frac{1}{a\,x+b\,x\,z} = \frac{1}{x\,(a+b\,z)}$
- **Rule:**

$$\int \frac{1}{a\,x + b\,x\,\text{Log}[c\,x^n]^m}\,dx \longrightarrow \int \frac{1}{x\,(a + b\,\text{Log}[c\,x^n]^m)}\,dx$$

- **Program code:**

```
Int[1/(a_.*x_+b_.*x_*Log[c_.*x_^n_.]^m_.),x_Symbol] :=
  Int[1/(x*(a+b*Log[c*x^n]^m)),x] /;
FreeQ[{a,b,c,m,n},x]
```

$$\int \text{Log}\left[c + d\, f^{a+b\,x}\right] dx$$

- **Derivation: Primitive rule**
- **Basis:** $\partial_x \text{PolyLog}[2, -c\, e^x] = -\text{Log}[1 + c\, e^x]$
- **Rule:**

$$\int \text{Log}\left[1 + c\, f^{a+b\,x}\right] dx \longrightarrow -\frac{\text{PolyLog}\left[2, -c\, f^{a+b\,x}\right]}{b\, \text{Log}[f]}$$

- **Program code:**

```
Int[Log[1+c_.*f_^(a_.+b_.*x_)],x_Symbol] :=
  -PolyLog[2,-c*f^(a+b*x)]/(b*Log[f]) /;
FreeQ[{a,b,c,f},x]
```

- **Derivation: Integration by parts**
- **Basis:** $\partial_x \text{Log}[c + d\, g[x]] = \partial_x \text{Log}\left[1 + \frac{d\, g[x]}{c}\right]$
- **Rule: If** $c \neq 1$**, then**

$$\int \text{Log}\left[c + d\, f^{a+b\,x}\right] dx \longrightarrow x\, \text{Log}\left[c + d\, f^{a+b\,x}\right] - x\, \text{Log}\left[1 + \frac{d\, f^{a+b\,x}}{c}\right] + \int \text{Log}\left[1 + \frac{d\, f^{a+b\,x}}{c}\right] dx$$

- **Program code:**

```
Int[Log[c_+d_.*f_^(a_.+b_.*x_)],x_Symbol] :=
  x*Log[c+d*f^(a+b*x)] - x*Log[1+d/c*f^(a+b*x)] +
  Int[Log[1+d/c*f^(a+b*x)],x] /;
FreeQ[{a,b,c,d,f},x] && NonzeroQ[c-1]
```

$$\int x^m \operatorname{Log}\left[c+d\, f^{a+b\,x}\right] dx$$

- **Derivation: Integration by parts**
- **Rule: If $m>0$, then**

$$\int x^m \operatorname{Log}\left[1+c\, f^{a+b\,x}\right] dx \longrightarrow -\frac{x^m \operatorname{PolyLog}\left[2,-c\, f^{a+b\,x}\right]}{b \operatorname{Log}[f]}+\frac{m}{b \operatorname{Log}[f]} \int x^{m-1} \operatorname{PolyLog}\left[2,-c\, f^{a+b\,x}\right] dx$$

- **Program code:**

```
Int[x_^m_.*Log[1+c_.*f_^(a_.+b_.*x_)],x_Symbol] :=
  -x^m*PolyLog[2,-c*f^(a+b*x)]/(b*Log[f]) +
  Dist[m/(b*Log[f]),Int[x^(m-1)*PolyLog[2,-c*f^(a+b*x)],x]] /;
FreeQ[{a,b,c,f},x] && RationalQ[m] && m>0
```

- **Derivation: Integration by parts**
- **Basis:** $\partial_x \operatorname{Log}[c+d\, g[x]] = \partial_x \operatorname{Log}\left[1+\frac{d\, g[x]}{c}\right]$
- **Rule: If $c \neq 1 \land m>0$, then**

$$\int x^m \operatorname{Log}\left[c+d\, f^{a+b\,x}\right] dx \longrightarrow \frac{x^{m+1} \operatorname{Log}\left[c+d\, f^{a+b\,x}\right]}{m+1}-\frac{x^{m+1} \operatorname{Log}\left[1+\frac{d\, f^{a+b\,x}}{c}\right]}{m+1}+\int x^m \operatorname{Log}\left[1+\frac{d\, f^{a+b\,x}}{c}\right] dx$$

- **Program code:**

```
Int[x_^m_.*Log[c_+d_.*f_^(a_.+b_.*x_)],x_Symbol] :=
  x^(m+1)*Log[c+d*f^(a+b*x)]/(m+1) - x^(m+1)*Log[1+d/c*f^(a+b*x)]/(m+1) +
  Int[x^m*Log[1+d/c*f^(a+b*x)],x] /;
FreeQ[{a,b,c,d,f},x] && NonzeroQ[c-1] && RationalQ[m] && m>0
```

$$\int \mathrm{Log}[u] \, dx$$

- **Reference: A&S 4.1.53**
- **Derivation: Integration by parts**
- **Rule: If u is an algebraic function of x, then**

$$\int \mathrm{Log}[u] \, dx \longrightarrow x \, \mathrm{Log}[u] - \int \frac{x \, \partial_x u}{u} \, dx$$

- **Program code:**

```
Int[Log[u_],x_Symbol] :=
  x*Log[u] -
  Int[Regularize[x*D[u,x]/u,x],x] /;
AlgebraicFunctionQ[u,x]
```