$$\int \left(f^{a+b\,x}\right)^p \mathrm{Sin}[c+d\,x]^n \, dx$$

- **Reference: CRC 533, A&S 4.3.136**
- **Rule: If** $d^2+b^2\,p^2\,\mathrm{Log}[f]^2\neq 0$**, then**

$$\int \left(f^{a+b\,x}\right)^p \mathrm{Sin}[c+d\,x]\, dx \longrightarrow \frac{b\,p\,\mathrm{Log}[f]\,\left(f^{a+b\,x}\right)^p \mathrm{Sin}[c+d\,x]}{d^2+b^2\,p^2\,\mathrm{Log}[f]^2}-\frac{d\,\left(f^{a+b\,x}\right)^p \mathrm{Cos}[c+d\,x]}{d^2+b^2\,p^2\,\mathrm{Log}[f]^2}$$

- **Program code:**

```
Int[(f_^(a_.+b_.*x_))^p_.*Sin[c_.+d_.*x_],x_Symbol] :=
  b*p*Log[f]*(f^(a+b*x))^p*Sin[c+d*x]/(d^2+b^2*p^2*Log[f]^2) -
  d*(f^(a+b*x))^p*Cos[c+d*x]/(d^2+b^2*p^2*Log[f]^2) /;
FreeQ[{a,b,c,d,f,p},x] && NonzeroQ[d^2+b^2*p^2*Log[f]^2]
```

- **Reference: CRC 538, A&S 4.3.137**

```
Int[(f_^(a_.+b_.*x_))^p_.*Cos[c_.+d_.*x_],x_Symbol] :=
  b*p*Log[f]*(f^(a+b*x))^p*Cos[c+d*x]/(d^2+b^2*p^2*Log[f]^2) +
  d*(f^(a+b*x))^p*Sin[c+d*x]/(d^2+b^2*p^2*Log[f]^2) /;
FreeQ[{a,b,c,d,f,p},x] && NonzeroQ[d^2+b^2*p^2*Log[f]^2]
```

- **Reference: CRC 542, A&S 4.3.138**
- **Rule: If** $d^2\,n^2+b^2\,p^2\,\mathrm{Log}[f]^2\neq 0\,\land\,n>1$**, then**

$$\int \left(f^{a+b\,x}\right)^p \mathrm{Sin}[c+d\,x]^n\, dx \longrightarrow \frac{b\,p\,\mathrm{Log}[f]\,\left(f^{a+b\,x}\right)^p \mathrm{Sin}[c+d\,x]^n}{d^2\,n^2+b^2\,p^2\,\mathrm{Log}[f]^2}-$$
$$\frac{d\,n\,\left(f^{a+b\,x}\right)^p \mathrm{Cos}[c+d\,x]\,\mathrm{Sin}[c+d\,x]^{n-1}}{d^2\,n^2+b^2\,p^2\,\mathrm{Log}[f]^2}+\frac{n\,(n-1)\,d^2}{d^2\,n^2+b^2\,p^2\,\mathrm{Log}[f]^2}\int \left(f^{a+b\,x}\right)^p \mathrm{Sin}[c+d\,x]^{n-2}\, dx$$

- **Program code:**

```
Int[(f_^(a_.+b_.*x_))^p_.*Sin[c_.+d_.*x_]^n_,x_Symbol] :=
  b*p*Log[f]*(f^(a+b*x))^p*Sin[c+d*x]^n/(d^2*n^2+b^2*p^2*Log[f]^2) -
  d*n*(f^(a+b*x))^p*Cos[c+d*x]*Sin[c+d*x]^(n-1)/(d^2*n^2+b^2*p^2*Log[f]^2) +
  Dist[(n*(n-1)*d^2)/(d^2*n^2+b^2*p^2*Log[f]^2),Int[(f^(a+b*x))^p*Sin[c+d*x]^(n-2),x]] /;
FreeQ[{a,b,c,d,f,p},x] && NonzeroQ[d^2*n^2+b^2*p^2*Log[f]^2] && RationalQ[n] && n>1
```

- **Reference: CRC 543, A&S 4.3.139**

```
Int[(f_^(a_.+b_.*x_))^p_.*Cos[c_.+d_.*x_]^m_,x_Symbol] :=
  b*p*Log[f]*(f^(a+b*x))^p*Cos[c+d*x]^m/(d^2*m^2+b^2*p^2*Log[f]^2) +
  d*m*(f^(a+b*x))^p*Sin[c+d*x]*Cos[c+d*x]^(m-1)/(d^2*m^2+b^2*p^2*Log[f]^2) +
  Dist[(m*(m-1)*d^2)/(d^2*m^2+b^2*p^2*Log[f]^2),Int[(f^(a+b*x))^p*Cos[c+d*x]^(m-2),x]] /;
FreeQ[{a,b,c,d,f,p},x] && NonzeroQ[d^2*m^2+b^2*p^2*Log[f]^2] && RationalQ[m] && m>1
```

- **Reference: CRC 551 when** $d^2\ (n+2)^2 + b^2\ p^2\ \mathrm{Log}[f]^2 = 0$

- **Rule: If** $d^2\ (n+2)^2 + b^2\ p^2\ \mathrm{Log}[f]^2 = 0 \wedge n+1 \neq 0 \wedge n+2 \neq 0$**, then**

$$\int \left(f^{a+b\,x}\right)^p \mathrm{Sin}[c+d\,x]^n\, dx \longrightarrow -\frac{b\,p\,\mathrm{Log}[f]\ \left(f^{a+b\,x}\right)^p \mathrm{Sin}[c+d\,x]^{n+2}}{d^2\ (n+1)\ (n+2)} + \frac{\left(f^{a+b\,x}\right)^p \mathrm{Cos}[c+d\,x]\ \mathrm{Sin}[c+d\,x]^{n+1}}{d\ (n+1)}$$

- **Program code:**

```
Int[(f_^(a_.+b_.*x_))^p_.*Sin[c_.+d_.*x_]^n_,x_Symbol] :=
  -b*p*Log[f]*(f^(a+b*x))^p*Sin[c+d*x]^(n+2)/(d^2*(n+1)*(n+2)) +
  (f^(a+b*x))^p*Cos[c+d*x]*Sin[c+d*x]^(n+1)/(d*(n+1)) /;
FreeQ[{a,b,c,d,f,n,p},x] && ZeroQ[d^2*(n+2)^2+b^2*p^2*Log[f]^2] && NonzeroQ[n+1] && NonzeroQ[n+2]
```

- **Reference: CRC 552 when** $d^2\ (n+2)^2 + b^2\ p^2\ \mathrm{Log}[f]^2 = 0$

```
Int[(f_^(a_.+b_.*x_))^p_.*Cos[c_.+d_.*x_]^n_,x_Symbol] :=
  -b*p*Log[f]*(f^(a+b*x))^p*Cos[c+d*x]^(n+2)/(d^2*(n+1)*(n+2)) -
  (f^(a+b*x))^p*Sin[c+d*x]*Cos[c+d*x]^(n+1)/(d*(n+1)) /;
FreeQ[{a,b,c,d,f,n,p},x] && ZeroQ[d^2*(n+2)^2+b^2*p^2*Log[f]^2] && NonzeroQ[n+1] && NonzeroQ[n+2]
```

- **Reference: CRC 551, CRC 542 inverted**

- **Rule: If** $d^2\ (n+2)^2 + b^2\ p^2\ \mathrm{Log}[f]^2 \neq 0 \wedge n < -1 \wedge n \neq -2$**, then**

$$\int \left(f^{a+b\,x}\right)^p \mathrm{Sin}[c+d\,x]^n\, dx \longrightarrow -\frac{b\,p\,\mathrm{Log}[f]\ \left(f^{a+b\,x}\right)^p \mathrm{Sin}[c+d\,x]^{n+2}}{d^2\ (n+1)\ (n+2)} +$$
$$\frac{\left(f^{a+b\,x}\right)^p \mathrm{Cos}[c+d\,x]\ \mathrm{Sin}[c+d\,x]^{n+1}}{d\ (n+1)} + \frac{d^2\ (n+2)^2 + b^2\ p^2\ \mathrm{Log}[f]^2}{d^2\ (n+1)\ (n+2)} \int \left(f^{a+b\,x}\right)^p \mathrm{Sin}[c+d\,x]^{n+2}\, dx$$

- **Program code:**

```
Int[(f_^(a_.+b_.*x_))^p_.*Sin[c_.+d_.*x_]^n_,x_Symbol] :=
  -b*p*Log[f]*(f^(a+b*x))^p*Sin[c+d*x]^(n+2)/(d^2*(n+1)*(n+2)) +
  (f^(a+b*x))^p*Cos[c+d*x]*Sin[c+d*x]^(n+1)/(d*(n+1)) +
  Dist[(d^2*(n+2)^2+b^2*p^2*Log[f]^2)/(d^2*(n+1)*(n+2)),Int[(f^(a+b*x))^p*Sin[c+d*x]^(n+2),x]] /;
FreeQ[{a,b,c,d,f,p},x] && NonzeroQ[d^2*(n+2)^2+b^2*p^2*Log[f]^2] && RationalQ[n] && n<-1 && n≠-2
```

- **Reference: CRC 552, CRC 543 inverted**

```
Int[(f_^(a_.+b_.*x_))^p_.*Cos[c_.+d_.*x_]^n_,x_Symbol] :=
  -b*p*Log[f]*(f^(a+b*x))^p*Cos[c+d*x]^(n+2)/(d^2*(n+1)*(n+2)) -
  (f^(a+b*x))^p*Sin[c+d*x]*Cos[c+d*x]^(n+1)/(d*(n+1)) +
  Dist[(d^2*(n+2)^2+b^2*p^2*Log[f]^2)/(d^2*(n+1)*(n+2)),Int[(f^(a+b*x))^p*Cos[c+d*x]^(n+2),x]] /;
FreeQ[{a,b,c,d,f,p},x] && NonzeroQ[d^2*(n+2)^2+b^2*p^2*Log[f]^2] && RationalQ[n] && n<-1 && n≠-2
```

$$\int \left(f^{a+b x}\right)^p \operatorname{Sec}[c+d x]^n \, dx$$

- **Reference: CRC 552 with** $b^2 p^2 \operatorname{Log}[f]^2 + d^2 (n-2)^2 = 0$

- **Rule: If** $b^2 p^2 \operatorname{Log}[f]^2 + d^2 (n-2)^2 = 0 \wedge n-1 \neq 0 \wedge n-2 \neq 0$**, then**

$$\int \left(f^{a+b x}\right)^p \operatorname{Sec}[c+d x]^n \, dx \longrightarrow -\frac{b\, p \operatorname{Log}[f] \left(f^{a+b x}\right)^p \operatorname{Sec}[c+d x]^{n-2}}{d^2 (n-1)(n-2)} + \frac{\left(f^{a+b x}\right)^p \operatorname{Sec}[c+d x]^{n-1} \operatorname{Sin}[c+d x]}{d (n-1)}$$

- **Program code:**

```
Int[(f_^(a_.+b_.*x_))^p_.*Sec[c_.+d_.*x_]^n_,x_Symbol] :=
  -b*p*Log[f]*(f^(a+b*x))^p*Sec[c+d*x]^(n-2)/(d^2*(n-1)*(n-2)) +
  (f^(a+b*x))^p*Sec[c+d*x]^(n-1)*Sin[c+d*x]/(d*(n-1)) /;
FreeQ[{a,b,c,d,f,p,n},x] && ZeroQ[b^2*p^2*Log[f]^2+d^2*(n-2)^2] && NonzeroQ[n-1] && NonzeroQ[n-2]
```

- **Reference: CRC 551 with** $b^2 p^2 \operatorname{Log}[f]^2 + d^2 (n-2)^2 = 0$

```
Int[(f_^(a_.+b_.*x_))^p_.*Csc[c_.+d_.*x_]^n_,x_Symbol] :=
  -b*p*Log[f]*(f^(a+b*x))^p*Csc[c+d*x]^(n-2)/(d^2*(n-1)*(n-2)) +
  (f^(a+b*x))^p*Csc[c+d*x]^(n-1)*Cos[c+d*x]/(d*(n-1)) /;
FreeQ[{a,b,c,d,f,p,n},x] && ZeroQ[b^2*p^2*Log[f]^2+d^2*(n-2)^2] && NonzeroQ[n-1] && NonzeroQ[n-2]
```

- **Reference: CRC 552**

- **Rule: If** $b^2 p^2 \operatorname{Log}[f]^2 + d^2 (n-2)^2 \neq 0 \wedge n > 1 \wedge n \neq 2$**, then**

$$\int \left(f^{a+b x}\right)^p \operatorname{Sec}[c+d x]^n \, dx \longrightarrow -\frac{b\, p \operatorname{Log}[f] \left(f^{a+b x}\right)^p \operatorname{Sec}[c+d x]^{n-2}}{d^2 (n-1)(n-2)} + \frac{\left(f^{a+b x}\right)^p \operatorname{Sec}[c+d x]^{n-1} \operatorname{Sin}[c+d x]}{d (n-1)} + \frac{b^2 p^2 \operatorname{Log}[f]^2 + d^2 (n-2)^2}{d^2 (n-1)(n-2)} \int \left(f^{a+b x}\right)^p \operatorname{Sec}[c+d x]^{n-2} \, dx$$

- **Program code:**

```
Int[(f_^(a_.+b_.*x_))^p_.*Sec[c_.+d_.*x_]^n_,x_Symbol] :=
  -b*p*Log[f]*(f^(a+b*x))^p*Sec[c+d*x]^(n-2)/(d^2*(n-1)*(n-2)) +
  (f^(a+b*x))^p*Sec[c+d*x]^(n-1)*Sin[c+d*x]/(d*(n-1)) +
  Dist[(b^2*p^2*Log[f]^2+d^2*(n-2)^2)/(d^2*(n-1)*(n-2)),Int[(f^(a+b*x))^p*Sec[c+d*x]^(n-2),x]] /;
FreeQ[{a,b,c,d,f,p},x] && NonzeroQ[b^2*p^2*Log[f]^2+d^2*(n-2)^2] && RationalQ[n] && n>1 && n≠2
```

- **Reference: CRC 551**

```
Int[(f_^(a_.+b_.*x_))^p_.*Csc[c_.+d_.*x_]^n_,x_Symbol] :=
  -b*p*Log[f]*(f^(a+b*x))^p*Csc[c+d*x]^(n-2)/(d^2*(n-1)*(n-2)) -
  (f^(a+b*x))^p*Csc[c+d*x]^(n-1)*Cos[c+d*x]/(d*(n-1)) +
  Dist[(b^2*p^2*Log[f]^2+d^2*(n-2)^2)/(d^2*(n-1)*(n-2)),Int[(f^(a+b*x))^p*Csc[c+d*x]^(n-2),x]] /;
FreeQ[{a,b,c,d,f,p},x] && NonzeroQ[b^2*p^2*Log[f]^2+d^2*(n-2)^2] && RationalQ[n] && n>1 && n≠2
```

$$\int x^m \left(f^{a+b\,x}\right)^p \text{Sin}[c+d\,x]^n \, dx$$

- **Derivation: Integration by parts**
- **Note: Each term of the sum $x^{m-1}\,u$ will be similar in form to the original integrand, but the degree of the monomial will be smaller by one.**
- **Rule: If $m>0 \land n\in\mathbb{Z} \land n>0$, let $u=\int\left(f^{a+b\,x}\right)^p \text{Sin}[c+d\,x]\,dx$, then**

$$\int x^m \left(f^{a+b\,x}\right)^p \text{Sin}[c+d\,x]^n \, dx \longrightarrow x^m\,u - m\int x^{m-1}\,u\,dx$$

- **Program code:**

```
Int[x_^m_.*(f_^(a_.+b_.*x_))^p_.*Sin[c_.+d_.*x_]^n_.,x_Symbol] :=
  Module[{u=Block[{ShowSteps=False,StepCounter=Null}, Int[(f^(a+b*x))^p*Sin[c+d*x]^n,x]]},
  x^m*u - Dist[m,Int[x^(m-1)*u,x]]] /;
FreeQ[{a,b,c,d,f,p},x] && RationalQ[m] && m>0 && IntegerQ[n] && n>0
```

```
Int[x_^m_.*(f_^(a_.+b_.*x_))^p_.*Cos[c_.+d_.*x_]^n_.,x_Symbol] :=
  Module[{u=Block[{ShowSteps=False,StepCounter=Null}, Int[(f^(a+b*x))^p*Cos[c+d*x]^n,x]]},
  x^m*u - Dist[m,Int[x^(m-1)*u,x]]] /;
FreeQ[{a,b,c,d,f,p},x] && RationalQ[m] && m>0 && IntegerQ[n] && n>0
```

$$\int f^v \operatorname{Sin}[w]^n \, dx$$

- **Derivation: Algebraic expansion**
- **Basis:** $\operatorname{Sin}[z] = \frac{i}{2 e^{i z}} - \frac{i e^{i z}}{2}$
- **Rule: If v and w are quadratic polynomials in x, then**

$$\int f^v \operatorname{Sin}[w] \, dx \longrightarrow \frac{i}{2} \int \frac{f^v}{e^{i w}} \, dx - \frac{i}{2} \int f^v e^{i w} \, dx$$

- **Program code:**

```
Int[f_^v_*Sin[w_],x_Symbol] :=
  Dist[I/2,Int[f^v/E^(I*w),x]] -
  Dist[I/2,Int[f^v*E^(I*w),x]] /;
FreeQ[f,x] && PolynomialQ[v,x] && Exponent[v,x]≤2 && PolynomialQ[w,x] && Exponent[w,x]≤2
```

- **Basis:** $\operatorname{Cos}[z] = \frac{e^{i z}}{2} + \frac{1}{2 e^{i z}}$

```
Int[f_^v_*Cos[w_],x_Symbol] :=
  Dist[1/2,Int[f^v*E^(I*w),x]] +
  Dist[1/2,Int[f^v/E^(I*w),x]] /;
FreeQ[f,x] && PolynomialQ[v,x] && Exponent[v,x]≤2 && PolynomialQ[w,x] && Exponent[w,x]≤2
```

- **Derivation: Algebraic expansion**
- **Basis:** $\operatorname{Sin}[z] = \frac{i}{2} \left(\frac{1}{e^{i z}} - e^{i z} \right)$
- **Rule: If $n \in \mathbb{Z} \land n > 0 \land$ v and w are quadratic polynomials in x, then**

$$\int f^v \operatorname{Sin}[w]^n \, dx \longrightarrow \left(\frac{i}{2} \right)^n \int f^v \left(\frac{1}{e^{i w}} - e^{i w} \right)^n dx$$

- **Program code:**

```
Int[f_^v_*Sin[w_]^n_,x_Symbol] :=
  Dist[(I/2)^n,Int[f^v*(1/E^(I*w)-E^(I*w))^n,x]] /;
FreeQ[f,x] && IntegerQ[n] && n>0 && PolynomialQ[v,x] && Exponent[v,x]≤2 &&
  PolynomialQ[w,x] && Exponent[w,x]≤2
```

- **Basis:** $\operatorname{Cos}[z] = \frac{1}{2} \left(e^{i z} + \frac{1}{e^{i z}} \right)$

```
Int[f_^v_*Cos[w_]^n_,x_Symbol] :=
  Dist[1/2^n,Int[f^v*(E^(I*w)+1/E^(I*w))^n,x]] /;
FreeQ[f,x] && IntegerQ[n] && n>0 && PolynomialQ[v,x] && Exponent[v,x]≤2 &&
  PolynomialQ[w,x] && Exponent[w,x]≤2
```